Rules for integrands of the form $(e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q$

0. $\int (e\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,dx$

   1. $\int (e\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $m\in\mathbb{Z}\,\lor\,e>0$

      1: $\int (e\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $(m\in\mathbb{Z}\,\lor\,e>0)\,\land\,\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Algebraic expansion and integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,(b\,x^n)^p == \frac{1}{b^{\frac{m+1}{n}-1}}\,x^{n-1}\,(b\,x^n)^{p+\frac{m+1}{n}-1}$

Basis: $x^{n-1}\,F[x^n] == \frac{1}{n}\,\text{Subst}[F[x],\,x,\,x^n]\,\partial_x x^n$

Rule 1.1.3.4.0.1.1: If $(m\in\mathbb{Z}\,\lor\,e>0)\,\land\,\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (e\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow\,\frac{e^m}{n\,b^{\frac{m+1}{n}-1}}\,\text{Subst}\Big[\int (b\,x)^{p+\frac{m+1}{n}-1}\,(c+d\,x)^q\,dx,\,x,\,x^n\Big]$$

Program code:

```
Int[(e_.*x_)^m_.*(b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  e^m/(n*b^(Simplify[(m+1)/n]-1))*Subst[Int[(b*x)^(p+Simplify[(m+1)/n]-1)*(c+d*x)^q,x],x,x^n] /;
FreeQ[{b,c,d,e,m,n,p,q},x] && (IntegerQ[m] || GtQ[e,0]) && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (e\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $(m\in\mathbb{Z}\,\lor\,e>0)\,\land\,\frac{m+1}{n}\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,x^n)^p}{x^{n\,p}} == 0$

Rule 1.1.3.4.0.1.2: If $(m\in\mathbb{Z}\,\lor\,e>0)\,\land\,\frac{m+1}{n}\notin\mathbb{Z}$, then

$$\int (e\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow\,\frac{e^m\,b^{\text{IntPart}[p]}\,(b\,x^n)^{\text{FracPart}[p]}}{x^{n\,\text{FracPart}[p]}}\int x^{m+n\,p}\,(c+d\,x^n)^q\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(b_.*x_^n_.)^p_*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  e^m*b^IntPart[p]*(b*x^n)^FracPart[p]/x^(n*FracPart[p])*Int[x^(m+n*p)*(c+d*x^n)^q,x] /;
FreeQ[{b,c,d,e,m,n,p,q},x] && (IntegerQ[m] || GtQ[e,0]) && Not[IntegerQ[Simplify[(m+1)/n]]]
```

2: $\int (e\,x)^m \left(b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Rule 1.1.3.4.0.2: If $m\notin\mathbb{Z}$, then

$$\int (e\,x)^m \left(b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$$

Program code:

```
Int[(e_*x_)^m_*(b_.*x_^n_.)^p_*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{b,c,d,e,m,n,p,q},x] && Not[IntegerQ[m]]
```

E1. $\int \frac{x^m}{(a+b x^2)^{1/4} (c+d x^2)} \, dx$ when $b c - 2 a d == 0 \land m \in \mathbb{Z} \land \left(a > 0 \lor \frac{m}{2} \in \mathbb{Z}\right)$

1: $\int \frac{x}{(a+b x^2)^{1/4} (c+d x^2)} \, dx$ when $b c - 2 a d == 0 \land a > 0$

Note: The result is real and continuous when the integrand is, and substitution $u \to x^2$ results in 2 inverse trig and 2 log terms.

Rule 1.1.3.4.E1.1: If $b c - 2 a d == 0 \land a > 0$, then

$$\int \frac{x}{(a+b x^2)^{1/4} (c+d x^2)} \, dx \to -\frac{1}{\sqrt{2}\, a^{1/4} d} \mathrm{ArcTan}\left[\frac{\sqrt{a}-\sqrt{a+b x^2}}{\sqrt{2}\, a^{1/4} (a+b x^2)^{1/4}}\right] - \frac{1}{\sqrt{2}\, a^{1/4} d} \mathrm{ArcTanh}\left[\frac{\sqrt{a}+\sqrt{a+b x^2}}{\sqrt{2}\, a^{1/4} (a+b x^2)^{1/4}}\right]$$

Program code:

```
Int[x_/((a_+b_.*x_^2)^(1/4)*(c_+d_.*x_^2)),x_Symbol] :=
  -1/(Sqrt[2]*Rt[a,4]*d)*ArcTan[(Rt[a,4]^2-Sqrt[a+b*x^2])/(Sqrt[2]*Rt[a,4]*(a+b*x^2)^(1/4))] -
  1/(Sqrt[2]*Rt[a,4]*d)*ArcTanh[(Rt[a,4]^2+Sqrt[a+b*x^2])/(Sqrt[2]*Rt[a,4]*(a+b*x^2)^(1/4))] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && PosQ[a]
```

2: $\int \frac{x^m}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} \, dx$ when $b c-2 a d=0 \land m \in \mathbb{Z} \land \left(a>0 \lor \frac{m}{2} \in \mathbb{Z}\right)$

Rule 1.1.3.4.E1.2: If $b c-2 a d=0 \land m \in \mathbb{Z} \land \left(a>0 \lor \frac{m}{2} \in \mathbb{Z}\right)$, then

$$\int \frac{x^m}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} \, dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{x^m}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)}, x\right] dx$$

Program code:

```
Int[x_^m_/((a_+b_.*x_^2)^(1/4)*(c_+d_.*x_^2)),x_Symbol] :=
  Int[ExpandIntegrand[x^m/((a+b*x^2)^(1/4)*(c+d*x^2)),x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && IntegerQ[m] && (PosQ[a] || IntegerQ[m/2])
```

E2. $\int \frac{x^m}{\left(a+b x^2\right)^{3/4} \left(c+d x^2\right)} \, dx$ when $b c-2 a d==0 \land m \in \mathbb{Z} \land \left(a>0 \lor \frac{m}{2} \in \mathbb{Z}\right)$

1. $\int \frac{x^2}{\left(a+b x^2\right)^{3/4} \left(c+d x^2\right)} \, dx$ when $b c-2 a d==0$

1: $\int \frac{x^2}{\left(a+b x^2\right)^{3/4} \left(c+d x^2\right)} \, dx$ when $b c-2 a d==0 \land \frac{b^2}{a}>0$

Reference: Eneström index number E688 in The Euler Archive

Rule 1.1.3.4.E2.1.1: If $b c-2 a d==0 \land \frac{b^2}{a}>0$, then

$$\int \frac{x^2}{\left(a+b x^2\right)^{3/4} \left(c+d x^2\right)} \, dx \to -\frac{b}{a d \left(\frac{b^2}{a}\right)^{3/4}} \operatorname{ArcTan}\left[\frac{b+\sqrt{\frac{b^2}{a}} \sqrt{a+b x^2}}{\left(\frac{b^2}{a}\right)^{3/4} x \left(a+b x^2\right)^{1/4}}\right]+\frac{b}{a d \left(\frac{b^2}{a}\right)^{3/4}} \operatorname{ArcTanh}\left[\frac{b-\sqrt{\frac{b^2}{a}} \sqrt{a+b x^2}}{\left(\frac{b^2}{a}\right)^{3/4} x \left(a+b x^2\right)^{1/4}}\right]$$

Program code:

```
Int[x_^2/((a_+b_.*x_^2)^(3/4)*(c_+d_.*x_^2)),x_Symbol] :=
  -b/(a*d*Rt[b^2/a,4]^3)*ArcTan[(b+Rt[b^2/a,4]^2*Sqrt[a+b*x^2])/(Rt[b^2/a,4]^3*x*(a+b*x^2)^(1/4))] +
  b/(a*d*Rt[b^2/a,4]^3)*ArcTanh[(b-Rt[b^2/a,4]^2*Sqrt[a+b*x^2])/(Rt[b^2/a,4]^3*x*(a+b*x^2)^(1/4))] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && PosQ[b^2/a]
```

2: $\int \frac{x^2}{(a+b x^2)^{3/4} (c+d x^2)} \, dx$ when $b c - 2 a d == 0 \land \frac{b^2}{a} \not> 0$

Reference: Eneström index number E688 in The Euler Archive

Derivation: Integration by substitution

Basis: If $b c - 2 a d == 0$, then $\frac{x^2}{(a+b x^2)^{3/4} (c+d x^2)} == \frac{2 b}{d} \text{Subst}\left[\frac{x^2}{4 a+b^2 x^4}, x, \frac{x}{(a+b x^2)^{1/4}}\right] \partial_x \frac{x}{(a+b x^2)^{1/4}}$

Rule 1.1.3.4.E2.1.2: If $b c - 2 a d == 0 \land \frac{b^2}{a} \not> 0$, then

$$\int \frac{x^2}{(a+b x^2)^{3/4} (c+d x^2)} \, dx \longrightarrow \frac{2 b}{d} \text{Subst}\left[\int \frac{x^2}{4 a+b^2 x^4} \, dx, x, \frac{x}{(a+b x^2)^{1/4}}\right]$$

$$\longrightarrow -\frac{b}{\sqrt{2} \, a d \left(-\frac{b^2}{a}\right)^{3/4}} \text{ArcTan}\left[\frac{\left(-\frac{b^2}{a}\right)^{1/4} x}{\sqrt{2} \, (a+b x^2)^{1/4}}\right] + \frac{b}{\sqrt{2} \, a d \left(-\frac{b^2}{a}\right)^{3/4}} \text{ArcTanh}\left[\frac{\left(-\frac{b^2}{a}\right)^{1/4} x}{\sqrt{2} \, (a+b x^2)^{1/4}}\right]$$

Program code:

```
Int[x_^2/((a_+b_.*x_^2)^(3/4)*(c_+d_.*x_^2)),x_Symbol] :=
  -b/(Sqrt[2]*a*d*Rt[-b^2/a,4]^3)*ArcTan[(Rt[-b^2/a,4]*x)/(Sqrt[2]*(a+b*x^2)^(1/4))] +
  b/(Sqrt[2]*a*d*Rt[-b^2/a,4]^3)*ArcTanh[(Rt[-b^2/a,4]*x)/(Sqrt[2]*(a+b*x^2)^(1/4))] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && NegQ[b^2/a]
```

2: $\int \frac{x^m}{(a+b x^2)^{3/4} (c+d x^2)} dx$ when $b c - 2 a d == 0 \land m \in \mathbb{Z} \land \left(a > 0 \lor \frac{m}{2} \in \mathbb{Z}\right)$

Rule 1.1.3.4.E2.2: If $b c - 2 a d == 0 \land m \in \mathbb{Z}$, then

$$\int \frac{x^m}{(a+b x^2)^{3/4} (c+d x^2)} dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{x^m}{(a+b x^2)^{3/4} (c+d x^2)}, x\right] dx$$

Program code:

```
Int[x_^m_/((a_+b_.*x_^2)^(3/4)*(c_+d_.*x_^2)),x_Symbol] :=
  Int[ExpandIntegrand[x^m/((a+b*x^2)^(3/4)*(c+d*x^2)),x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && IntegerQ[m] && (PosQ[a] || IntegerQ[m/2])
```

1: $\int x^m (a+b x^n)^p (c+d x^n)^q dx$ when $b c - a d \neq 0 \land m - n + 1 == 0$

Derivation: Integration by substitution

Basis: $x^{n-1} F[x^n] == \frac{1}{n} \text{Subst}[F[x], x, x^n] \partial_x x^n$

Rule 1.1.3.4.1: If $b c - a d \neq 0 \land m - n + 1 == 0$, then

$$\int x^m (a+b x^n)^p (c+d x^n)^q dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int (a+b x)^p (c+d x)^q dx, x, x^n\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  1/n*Subst[Int[(a+b*x)^p*(c+d*x)^q,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && NeQ[b*c-a*d,0] && EqQ[m-n+1,0]
```

2: $\int x^m (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land (p \mid q) \in \mathbb{Z} \land n < 0$

Derivation: Algebraic expansion

Basis: If $p \in \mathbb{Z}$, then $(a+b x^n)^p == x^{n p} (b+a x^{-n})^p$

Rule 1.1.3.4.2: If $b c - a d \neq 0 \land (p \mid q) \in \mathbb{Z} \land n < 0$, then

$$\int x^m (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow \int x^{m+n (p+q)} (b+a x^{-n})^p (d+c x^{-n})^q \, dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  Int[x^(m+n*(p+q))*(b+a*x^(-n))^p*(d+c*x^(-n))^q,x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && IntegersQ[p,q] && NegQ[n]
```

3. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \, dx$ when $b\,c-a\,d\neq 0 \land \frac{m+1}{n}\in\mathbb{Z}$

1: $\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \, dx$ when $b\,c-a\,d\neq 0 \land \frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$

Note: If $n\in\mathbb{Z} \land \frac{m+1}{n}\in\mathbb{Z}$, then $m\in\mathbb{Z}$, and $(e\,x)^m$ automatically evaluates to $e^m\,x^m$.

Rule 1.1.3.4.3.1: If $b\,c-a\,d\neq 0 \land \frac{m+1}{n}\in\mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \, dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,(a+b\,x)^p\,(c+d\,x)^q \, dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*x)^p*(c+d*x)^q,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && NeQ[b*c-a*d,0] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0\ \land\ \frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Basis: $\frac{(e\,x)^m}{x^m} == \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.4.3.2: If $b\,c-a\,d\neq 0\ \land\ \frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$$

Program code:

```
Int[(e_*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && NeQ[b*c-a*d,0] && IntegerQ[Simplify[(m+1)/n]]
```

4: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\land\,(p\mid q)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.4.4: If $b\,c-a\,d\neq 0\,\land\,(p\mid q)\in\mathbb{Z}^+$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[(e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q,\,x\right]dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  Int[ExpandIntegrand[(e*x)^m*(a+b*x^n)^p*(c+d*x^n)^q,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b*c-a*d,0] && IGtQ[p,0] && IGtQ[q,0]
```

5. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq 0$

1: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)==0\,\land\,m\neq -1$

Derivation: Trinomial recurrence 2b with $c=0$ and $a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)==0$

Rule 1.1.3.4.5.1: If $b\,c-a\,d\neq 0\,\land\,a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)==0\,\land\,m\neq -1$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx\;\longrightarrow\;\frac{c\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}}{a\,e\,(m+1)}$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(a*e*(m+1)) /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[b*c-a*d,0] && EqQ[a*d*(m+1)-b*c*(m+n*(p+1)+1),0] && NeQ[m,-1]
```

2. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n\,(p+1)+1==0$

1: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n\,(p+1)+1==0\,\land\,(n\in\mathbb{Z}\,\lor\,e>0)\,\land\,(n>0\,\land\,m<-1\,\lor\,n<0\,\land\,m+n>-1)$

Derivation: Trinomial recurrence 3b with $c=0$

Rule 1.1.3.4.5.2.1: If $b\,c-a\,d\neq 0\,\land\,(n\in\mathbb{Z}\,\lor\,e>0)\,\land\,(n>0\,\land\,m<-1\,\lor\,n<0\,\land\,m+n>-1)$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx\,\longrightarrow\,\frac{c\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}}{a\,e\,(m+1)}+\frac{d}{e^n}\int (e\,x)^{m+n}\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(a*e*(m+1)) + d/e^n*Int[(e*x)^(m+n)*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[b*c-a*d,0] && EqQ[m+n*(p+1)+1,0] && (IntegerQ[n] || GtQ[e,0]) &&
  (GtQ[n,0] && LtQ[m,-1] || LtQ[n,0] && GtQ[m+n,-1])
```

2: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n\,(p+1)+1==0\,\land\,m\neq -1$

Derivation: Trinomial recurrence 2b with $c=0$

Rule 1.1.3.4.5.2.2: If $b\,c-a\,d\neq 0\,\land\,m+n\,(p+1)+1==0\,\land\,m\neq -1$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx\,\longrightarrow\,\frac{(b\,c-a\,d)\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}}{a\,b\,e\,(m+1)}+\frac{d}{b}\int (e\,x)^m\,(a+b\,x^n)^{p+1}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  (b*c-a*d)*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(a*b*e*(m+1)) + d/b*Int[(e*x)^m*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[b*c-a*d,0] && EqQ[m+n*(p+1)+1,0] && NeQ[m,-1]
```

3: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq 0\wedge(n\in\mathbb{Z}\vee e>0)\wedge(n>0\wedge m<-1\vee n<0\wedge m+n>-1)$

Derivation: Trinomial recurrence 3b with $c=0$

Rule 1.1.3.4.5.3: If $b\,c-a\,d\neq 0\wedge(n\in\mathbb{Z}\vee e>0)\wedge(n>0\wedge m<-1\vee n<0\wedge m+n>-1)$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx\longrightarrow\frac{c\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}}{a\,e\,(m+1)}+\frac{a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)}{a\,e^n\,(m+1)}\int (e\,x)^{m+n}\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(a*e*(m+1)) +
  (a*d*(m+1)-b*c*(m+n*(p+1)+1))/(a*e^n*(m+1))*Int[(e*x)^(m+n)*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b*c-a*d,0] && (IntegerQ[n] || GtQ[e,0]) &&
  (GtQ[n,0] && LtQ[m,-1] || LtQ[n,0] && GtQ[m+n,-1]) && Not[ILtQ[p,-1]]
```

4. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq 0\wedge p<-1$

1. $\int x^m\,(a+b\,x^2)^p\,(c+d\,x^2)\,dx$ when $b\,c-a\,d\neq 0\wedge p<-1\wedge\frac{m}{2}\in\mathbb{Z}\wedge(p\in\mathbb{Z}\vee m+2\,p+1==0)$

1: $\int x^m\,(a+b\,x^2)^p\,(c+d\,x^2)\,dx$ when $b\,c-a\,d\neq 0\wedge p<-1\wedge\frac{m}{2}\in\mathbb{Z}^+\wedge(p\in\mathbb{Z}\vee m+2\,p+1==0)$

Derivation: ???

Note: If $\frac{m}{2}\in\mathbb{Z}^+$, $b^{m/2}\,x^{m-2}\,(c+d\,x^2)-(-a)^{m/2-1}\,(b\,c-a\,d)$ is divisible by $a+b\,x^2$.

Note: The degree of the polynomial in the resulting integrand is m.

Note: This rule should be generalized for integrands of the form $x^m\,(a+b\,x^n)^p\,(c+d\,x^n)$.

Rule 1.1.3.4.5.4.1.1: If $b\,c-a\,d\neq 0\wedge p<-1\wedge\frac{m}{2}\in\mathbb{Z}^+\wedge(p\in\mathbb{Z}\vee m+2\,p+1==0)$, then

$$\int x^m \left(a+b x^2\right)^p \left(c+d x^2\right) dx \longrightarrow$$

$$\frac{(-a)^{m/2-1} \left(b c-a d\right) x \left(a+b x^2\right)^{p+1}}{2 b^{m/2+1} (p+1)}+$$

$$\frac{1}{2 b^{m/2+1} (p+1)} \int \left(a+b x^2\right)^{p+1} \left(\frac{2 b (p+1) x^2 \left(b^{m/2} x^{m-2} \left(c+d x^2\right)-(-a)^{m/2-1} \left(b c-a d\right)\right)}{a+b x^2}-(-a)^{m/2-1} \left(b c-a d\right)\right) dx$$

Program code:

```
Int[x_^m_*(a_+b_.*x_^2)^p_*(c_+d_.*x_^2),x_Symbol] :=
  (-a)^(m/2-1)*(b*c-a*d)*x*(a+b*x^2)^(p+1)/(2*b^(m/2+1)*(p+1)) +
  1/(2*b^(m/2+1)*(p+1))*Int[(a+b*x^2)^(p+1)*
    ExpandToSum[2*b*(p+1)*x^2*Together[(b^(m/2)*x^(m-2)*(c+d*x^2)-(-a)^(m/2-1)*(b*c-a*d))/(a+b*x^2)]-(-a)^(m/2-1)*(b*c-a*d),x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && IGtQ[m/2,0] && (IntegerQ[p] || EqQ[m+2*p+1,0])
```

2: $\int x^m \left(a+b x^2\right)^p \left(c+d x^2\right) dx$ when $b c-a d\neq 0\ \land\ p<-1\ \land\ \frac{m}{2}\in\mathbb{Z}^-\ \land\ (p\in\mathbb{Z}\ \lor\ m+2 p+1==0)$

Derivation: ???

Note: If $\frac{m}{2}\in\mathbb{Z}^-$, $b^{m/2}\left(c+d x^2\right)-(-a)^{m/2-1}(b c-a d)\, x^{-m+2}$ is divisible by $a+b x^2$.

Note: The degree of the polynomial in the resulting integrand is $-m$.

Note: This rule should be generalized for integrands of the form $x^m\left(a+b x^n\right)^p\left(c+d x^n\right)$.

Rule 1.1.3.4.5.4.1.2: If $b c-a d\neq 0\ \land\ p<-1\ \land\ \frac{m}{2}\in\mathbb{Z}^-\ \land\ (p\in\mathbb{Z}\ \lor\ m+2 p+1==0)$, then

$$\int x^m\left(a+b x^2\right)^p\left(c+d x^2\right) dx\ \longrightarrow$$
$$\frac{(-a)^{m/2-1}(b c-a d)\, x\left(a+b x^2\right)^{p+1}}{2 b^{m/2+1}(p+1)}+$$
$$\frac{1}{2 b^{m/2+1}(p+1)}\int x^m\left(a+b x^2\right)^{p+1}\left(\frac{2 b (p+1)\left(b^{m/2}\left(c+d x^2\right)-(-a)^{m/2-1}(b c-a d)\, x^{-m+2}\right)}{a+b x^2}-(-a)^{m/2-1}(b c-a d)\, x^{-m}\right) dx$$

Program code:

```
Int[x_^m_*(a_+b_.*x_^2)^p_*(c_+d_.*x_^2),x_Symbol] :=
  (-a)^(m/2-1)*(b*c-a*d)*x*(a+b*x^2)^(p+1)/(2*b^(m/2+1)*(p+1)) +
  1/(2*b^(m/2+1)*(p+1))*Int[x^m*(a+b*x^2)^(p+1)*
    ExpandToSum[2*b*(p+1)*Together[(b^(m/2)*(c+d*x^2)-(-a)^(m/2-1)*(b*c-a*d)*x^(-m+2))/(a+b*x^2)]-
      (-a)^(m/2-1)*(b*c-a*d)*x^(-m),x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && ILtQ[m/2,0] && (IntegerQ[p] || EqQ[m+2*p+1,0])
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx$ when $b\,c-a\,d\neq 0\ \land\ p<-1$

Derivation: Trinomial recurrence 2b with $c=0$

Rule 1.1.3.4.5.4.2: If $b\,c-a\,d\neq 0\ \land\ p<-1$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx \longrightarrow -\frac{\left(b\,c-a\,d\right)\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,b\,e\,n\,(p+1)}-\frac{a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)}{a\,b\,n\,(p+1)}\int (e\,x)^m \left(a+b\,x^n\right)^{p+1} dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  -(b*c-a*d)*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(a*b*e*n*(p+1)) -
  (a*d*(m+1)-b*c*(m+n*(p+1)+1))/(a*b*n*(p+1))*Int[(e*x)^m*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] &&
  (Not[IntegerQ[p+1/2]] && NeQ[p,-5/4] || Not[RationalQ[m]] || IGtQ[n,0] && ILtQ[p+1/2,0] && LeQ[-1,m,-n*(p+1)])
```

5: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx$ when $b\,c-a\,d\neq 0\ \land\ m+n\,(p+1)+1\neq 0$

Derivation: Trinomial recurrence 2b with $c=0$ composed with binomial recurrence 1b

Rule 1.1.3.4.5.5: If $b\,c-a\,d\neq 0\ \land\ m+n\,(p+1)+1\neq 0$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx \longrightarrow \frac{d\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{b\,e\,(m+n\,(p+1)+1)}-\frac{a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)}{b\,(m+n\,(p+1)+1)}\int (e\,x)^m \left(a+b\,x^n\right)^p dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  d*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(b*e*(m+n*(p+1)+1)) -
  (a*d*(m+1)-b*c*(m+n*(p+1)+1))/(b*(m+n*(p+1)+1))*Int[(e*x)^m*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[b*c-a*d,0] && NeQ[m+n*(p+1)+1,0]
```

6. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}$

1. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+$

0: $\int \frac{(e\,x)^m \left(a+b\,x^n\right)^p}{c+d\,x^n}\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.4.6.1.0: If $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p\in\mathbb{Z}^+$, then

$$\int \frac{(e\,x)^m \left(a+b\,x^n\right)^p}{c+d\,x^n}\,dx \,\longrightarrow\, \int \text{ExpandIntegrand}\left[\frac{(e\,x)^m \left(a+b\,x^n\right)^p}{c+d\,x^n},\,x\right]dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_/(c_+d_.*x_^n_),x_Symbol] :=
  Int[ExpandIntegrand[(e*x)^m*(a+b*x^n)^p/(c+d*x^n),x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && IGtQ[p,0] && (IntegerQ[m] || IGtQ[2*(m+1),0] || Not[RationalQ[m]])
```

1. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^2\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+$

   1: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^2\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,m<-1\,\land\,n>0$

Derivation: ?

Rule 1.1.3.4.6.1.1.1: If $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,m<-1\,\land\,n>0$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^2\,dx\,\longrightarrow$$
$$\frac{c^2\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}}{a\,e\,(m+1)}-\frac{1}{a\,e^n\,(m+1)}\int (e\,x)^{m+n}\,(a+b\,x^n)^p\,\left(b\,c^2\,n\,(p+1)+c\,(b\,c-2\,a\,d)\,(m+1)-a\,(m+1)\,d^2\,x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^2,x_Symbol] :=
  c^2*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(a*e*(m+1)) -
  1/(a*e^n*(m+1))*Int[(e*x)^(m+n)*(a+b*x^n)^p*Simp[b*c^2*n*(p+1)+c*(b*c-2*a*d)*(m+1)-a*(m+1)*d^2*x^n,x],x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[m,-1] && GtQ[n,0]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^2 dx$ when $b\,c-a\,d\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p<-1$

Derivation: ?

Rule 1.1.3.4.6.1.1.2: If $b\,c-a\,d\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p<-1$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^2 dx \longrightarrow$$

$$-\frac{(b\,c-a\,d)^2\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,b^2\,e\,n\,(p+1)}+\frac{1}{a\,b^2\,n\,(p+1)}\int (e\,x)^m \left(a+b\,x^n\right)^{p+1}\left((b\,c-a\,d)^2\,(m+1)+b^2\,c^2\,n\,(p+1)+a\,b\,d^2\,n\,(p+1)\,x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^2,x_Symbol] :=
  -(b*c-a*d)^2*(e*x)^(m+1)*(a+b*x^n)^(p+1)/(a*b^2*e*n*(p+1)) +
  1/(a*b^2*n*(p+1))*Int[(e*x)^m*(a+b*x^n)^(p+1)*Simp[(b*c-a*d)^2*(m+1)+b^2*c^2*n*(p+1)+a*b*d^2*n*(p+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[p,-1]
```

3: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^2 dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,m+n\,(p+2)+1\neq 0$

Derivation: ?

Rule 1.1.3.4.6.1.1.3: If $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,m+n\,(p+2)+1\neq 0$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^2 dx\,\longrightarrow$$
$$\frac{d^2\,(e\,x)^{m+n+1}\left(a+b\,x^n\right)^{p+1}}{b\,e^{n+1}\,(m+n\,(p+2)+1)}+$$
$$\frac{1}{b\,(m+n\,(p+2)+1)}\int (e\,x)^m \left(a+b\,x^n\right)^p \left(b\,c^2\,(m+n\,(p+2)+1)+d\,\left((2\,b\,c-a\,d)\,(m+n+1)+2\,b\,c\,n\,(p+1)\right)x^n\right) dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^2,x_Symbol] :=
  d^2*(e*x)^(m+n+1)*(a+b*x^n)^(p+1)/(b*e^(n+1)*(m+n*(p+2)+1)) +
  1/(b*(m+n*(p+2)+1))*Int[(e*x)^m*(a+b*x^n)^p*Simp[b*c^2*(m+n*(p+2)+1)+d*((2*b*c-a*d)*(m+n+1)+2*b*c*n*(p+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && NeQ[m+n*(p+2)+1,0]
```

2: $\int x^m \left(a+b x^n\right)^p \left(c+d x^n\right)^q dx$ when $b c-a d \neq 0 \wedge n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z} \wedge \text{GCD}[m+1, n] \neq 1$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z} \wedge m \in \mathbb{Z}$, let $k = \text{GCD}[m+1, n]$, then $x^m F[x^n] = \frac{1}{k} \text{Subst}\left[x^{\frac{m+1}{k}-1} F\left[x^{n/k}\right], x, x^k\right] \partial_x x^k$

Rule 1.1.3.4.6.1.2: If $b c-a d \neq 0 \wedge n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$, let $k = \text{GCD}[m+1, n]$, if $k \neq 1$, then

$$\int x^m \left(a+b x^n\right)^p \left(c+d x^n\right)^q dx \rightarrow \frac{1}{k} \text{Subst}\left[\int x^{\frac{m+1}{k}-1} \left(a+b x^{n/k}\right)^p \left(c+d x^{n/k}\right)^q dx, x, x^k\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  With[{k=GCD[m+1,n]},
  1/k*Subst[Int[x^((m+1)/k-1)*(a+b*x^(n/k))^p*(c+d*x^(n/k))^q,x],x,x^k] /;
 k≠1] /;
FreeQ[{a,b,c,d,p,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && IntegerQ[m]
```

3: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $(e\,x)^m\,F[x] = \frac{k}{e}\,\text{Subst}\!\left[x^{k\,(m+1)-1}\,F\!\left[\frac{x^k}{e}\right],\,x,\,(e\,x)^{1/k}\right]\,\partial_x\,(e\,x)^{1/k}$

Rule 1.1.3.4.6.1.3: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,m\in\mathbb{F}$, let $k=\text{Denominator}[m]$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow\;\frac{k}{e}\,\text{Subst}\!\left[\int x^{k\,(m+1)-1}\left(a+\frac{b\,x^{k\,n}}{e^n}\right)^p\left(c+\frac{d\,x^{k\,n}}{e^n}\right)^q dx,\,x,\,(e\,x)^{1/k}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/e*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(k*n)/e^n)^p*(c+d*x^(k*n)/e^n)^q,x],x,(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,p,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && FractionQ[m] && IntegerQ[p]
```

4. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1$

1. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,q>0$

1: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,q>0\,\wedge\,m-n+1>0$

Derivation: Binomial product recurrence 1 with $A=0$, $B=1$ and $m=m-n$

Derivation: Binomial product recurrence 3a with $A=c$, $B=d$ and $q=q-1$

Rule 1.1.3.4.6.1.4.1.1: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,q>0\,\wedge\,m-n+1>0$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow$$

$$\frac{e^{n-1}\,(e\,x)^{m-n+1}\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^q}{b\,n\,(p+1)}-\frac{e^n}{b\,n\,(p+1)}\int(e\,x)^{m-n}\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^{q-1}\,\left(c\,(m-n+1)+d\,(m+n\,(q-1)+1)\,x^n\right)\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  e^(n-1)*(e*x)^(m-n+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(b*n*(p+1)) -
  e^n/(b*n*(p+1))*Int[(e*x)^(m-n)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*Simp[c*(m-n+1)+d*(m+n*(q-1)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[p,-1] && GtQ[q,0] && GtQ[m-n+1,0] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2: $\int(e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,q>1$

Derivation: Binomial product recurrence 1 with $A = c$, $B = d$ and $q = q - 1$

Rule 1.1.3.4.6.1.4.1.2: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,q>1$, then

$$\int(e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,dx\longrightarrow$$

$$-\frac{\left(c\,b-a\,d\right)\,(e\,x)^{m+1}\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^{q-1}}{a\,b\,e\,n\,(p+1)}+\frac{1}{a\,b\,n\,(p+1)}\cdot$$

$$\int(e\,x)^m\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^{q-2}\,\left(c\,\left(c\,b\,n\,(p+1)+\left(c\,b-a\,d\right)\,(m+1)\right)+d\,\left(c\,b\,n\,(p+1)+\left(c\,b-a\,d\right)\,(m+n\,(q-1)+1)\right)\,x^n\right)\,dx$$

Programcode:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -(c*b-a*d)*(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)/(a*b*e*n*(p+1)) +
  1/(a*b*n*(p+1))*Int[(e*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-2)*
    Simp[c*(c*b*n*(p+1)+(c*b-a*d)*(m+1))+d*(c*b*n*(p+1)+(c*b-a*d)*(m+n*(q-1)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[p,-1] && GtQ[q,1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

3: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p<-1 \wedge 0<q<1$

Derivation: Binomial product recurrence 1 with $A = 1$ and $B = 0$

Derivation: Binomial product recurrence 3b with $A = c$, $B = d$ and $q = q-1$

Rule 1.1.3.4.6.1.4.1.3: If $b\,c-a\,d\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p<-1 \wedge 0<q<1$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx \longrightarrow$$
$$-\frac{(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^q}{a\,e\,n\,(p+1)}+$$
$$\frac{1}{a\,n\,(p+1)}\int (e\,x)^m \left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q-1}\left(c\,(m+n\,(p+1)+1)+d\,(m+n\,(p+q+1)+1)\,x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*e*n*(p+1)) +
  1/(a*n*(p+1))*Int[(e*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*Simp[c*(m+n*(p+1)+1)+d*(m+n*(p+q+1)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[p,-1] && LtQ[0,q,1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p<-1 \wedge m-n+1>0$

1: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p<-1 \wedge m-n+1>n$

Derivation: Binomial product recurrence 3a with $A = 0$, $B = 1$ and $m = m-n$

Rule 1.1.3.4.6.1.4.2.1: If $b\,c-a\,d\neq 0 \wedge n\in\mathbb{Z}^+ \wedge p<-1 \wedge m-n+1>n$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx \longrightarrow$$
$$-\frac{a\,e^{2\,n-1}\,(e\,x)^{m-2\,n+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q+1}}{b\,n\,(b\,c-a\,d)\,(p+1)}+$$

$$\frac{e^{2n}}{b\,n\,(b\,c-a\,d)\,(p+1)}\int (e\,x)^{m-2n}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q\,\left(a\,c\,(m-2\,n+1)+\left(a\,d\,(m-n+n\,q+1)+b\,c\,n\,(p+1)\right)x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -a*e^(2*n-1)*(e*x)^(m-2*n+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(b*n*(b*c-a*d)*(p+1)) +
  e^(2*n)/(b*n*(b*c-a*d)*(p+1))*Int[(e*x)^(m-2*n)*(a+b*x^n)^(p+1)*(c+d*x^n)^q*
    Simp[a*c*(m-2*n+1)+(a*d*(m-n+n*q+1)+b*c*n*(p+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m-n+1,n] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p<-1\,\land\,n\geq m-n+1>0$

Derivation: Binomial product recurrence 3a with $A=1$ and $B=0$

Derivation: Binomial product recurrence 3b with $A=0$, $B=1$ and $m=m-n$

Rule 1.1.3.4.6.1.4.2.2: If $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p<-1\,\land\,n\geq m-n+1>0$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow$$

$$\frac{e^{n-1}\,(e\,x)^{m-n+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^{q+1}}{n\,(b\,c-a\,d)\,(p+1)}-$$

$$\frac{e^n}{n\,(b\,c-a\,d)\,(p+1)}\int (e\,x)^{m-n}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q\,\left(c\,(m-n+1)+d\,(m+n\,(p+q+1)+1)\,x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  e^(n-1)*(e*x)^(m-n+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(n*(b*c-a*d)*(p+1)) -
  e^n/(n*(b*c-a*d)*(p+1))*Int[(e*x)^(m-n)*(a+b*x^n)^(p+1)*(c+d*x^n)^q*Simp[c*(m-n+1)+d*(m+n*(p+q+1)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[p,-1] && GeQ[n,m-n+1] && GtQ[m-n+1,0] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x
```

3: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p<-1$

Derivation: Binomial product recurrence 3b with $A=1$ and $B=0$

Rule 1.1.3.4.6.1.4.3: If $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,p<-1$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow$$
$$-\frac{b\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^{q+1}}{a\,e\,n\,(b\,c-a\,d)\,(p+1)}+$$
$$\frac{1}{a\,n\,(b\,c-a\,d)\,(p+1)}\int (e\,x)^m\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q\,\left(c\,b\,(m+1)+n\,(b\,c-a\,d)\,(p+1)+d\,b\,(m+n\,(p+q+2)+1)\,x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -b*(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*e*n*(b*c-a*d)*(p+1)) +
  1/(a*n*(b*c-a*d)*(p+1))*
    Int[(e*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^q*Simp[c*b*(m+1)+n*(b*c-a*d)*(p+1)+d*b*(m+n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[p,-1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

5. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,q>0$

1. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,q>0\,\land\,m<-1$

1: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,q>0\,\land\,m<-1\,\land\,p>0$

Derivation: Binomial product recurrence 2a with $A=a$, $B=b$ and $p=p-1$

Rule 1.1.3.4.6.1.5.1.1: If $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{Z}^+\land\,q>0\,\land\,m<-1\,\land\,p>0$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow$$

$$\frac{(e\,x)^{m+1}\,(a+b\,x^n)^p\,(c+d\,x^n)^q}{e\,(m+1)}-\frac{n}{e^n\,(m+1)}\int(e\,x)^{m+n}\,(a+b\,x^n)^{p-1}\,(c+d\,x^n)^{q-1}\,(b\,c\,p+a\,d\,q+b\,d\,(p+q)\,x^n)\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  (e*x)^(m+1)*(a+b*x^n)^p*(c+d*x^n)^q/(e*(m+1)) -
  n/(e^n*(m+1))*Int[(e*x)^(m+n)*(a+b*x^n)^(p-1)*(c+d*x^n)^(q-1)*Simp[b*c*p+a*d*q+b*d*(p+q)*x^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && GtQ[q,0] && LtQ[m,-1] && GtQ[p,0] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2: $\int(e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq0\,\land\,n\in\mathbb{Z}^+\land\,q>1\,\land\,m<-1$

Derivation: Binomial product recurrence 2a with $A=c$, $B=d$ and $q=q-1$

Rule 1.1.3.4.6.1.5.1.2: If $b\,c-a\,d\neq0\,\land\,n\in\mathbb{Z}^+\land\,q>1\,\land\,m<-1$, then

$$\int(e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow$$

$$\frac{c\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^{q-1}}{a\,e\,(m+1)}-\frac{1}{a\,e^n\,(m+1)}\cdot$$

$$\int(e\,x)^{m+n}\,(a+b\,x^n)^p\,(c+d\,x^n)^{q-2}\,\left(c\,(c\,b-a\,d)\,(m+1)+c\,n\,(b\,c\,(p+1)+a\,d\,(q-1))+d\,((c\,b-a\,d)\,(m+1)+c\,b\,n\,(p+q))\,x^n\right)\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  c*(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)/(a*e*(m+1)) -
  1/(a*e^n*(m+1))*Int[(e*x)^(m+n)*(a+b*x^n)^p*(c+d*x^n)^(q-2)*
    Simp[c*(c*b-a*d)*(m+1)+c*n*(b*c*(p+1)+a*d*(q-1))+d*((c*b-a*d)*(m+1)+c*b*n*(p+q))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && GtQ[q,1] && LtQ[m,-1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

3: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,0<q<1\,\wedge\,m<-1$

Derivation: Binomial product recurrence 2a with $A=1$ and $B=0$

Derivation: Binomial product recurrence 4b with $A=c$, $B=d$ and $q=q-1$

Rule 1.1.3.4.6.1.5.1.3: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,0<q<1\,\wedge\,m<-1$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow$$
$$\frac{(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q}{a\,e\,(m+1)}-$$
$$\frac{1}{a\,e^n\,(m+1)}\int (e\,x)^{m+n}\,(a+b\,x^n)^p\,(c+d\,x^n)^{q-1}\,\left(c\,b\,(m+1)+n\,(b\,c\,(p+1)+a\,d\,q)+d\,(b\,(m+1)+b\,n\,(p+q+1))\,x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  (e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*e*(m+1)) -
  1/(a*e^n*(m+1))*Int[(e*x)^(m+n)*(a+b*x^n)^p*(c+d*x^n)^(q-1)*
    Simp[c*b*(m+1)+n*(b*c*(p+1)+a*d*q)+d*(b*(m+1)+b*n*(p+q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[0,q,1] && LtQ[m,-1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q>0\,\wedge\,p>0$

Derivation: Binomial product recurrence 2b with $A=a$, $B=b$ and $p=p-1$

Rule 1.1.3.4.6.1.5.2: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q>0\,\wedge\,p>0$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow$$
$$\frac{(e\,x)^{m+1}\,(a+b\,x^n)^p\,(c+d\,x^n)^q}{e\,(m+n\,(p+q)+1)}+$$

$$\frac{n}{m+n\,(p+q)+1}\int (e\,x)^m\,\left(a+b\,x^n\right)^{p-1}\,\left(c+d\,x^n\right)^{q-1}\,\left(a\,c\,(p+q)+\left(q\,\left(b\,c-a\,d\right)+a\,d\,(p+q)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  (e*x)^(m+1)*(a+b*x^n)^p*(c+d*x^n)^q/(e*(m+n*(p+q)+1)) +
  n/(m+n*(p+q)+1)*Int[(e*x)^m*(a+b*x^n)^(p-1)*(c+d*x^n)^(q-1)*Simp[a*c*(p+q)+(q*(b*c-a*d)+a*d*(p+q))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && GtQ[q,0] && GtQ[p,0] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

3: $\int (e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q>1$

Derivation: Binomial product recurrence 2b with A = c, B = d and q = q − 1

Rule 1.1.3.4.6.1.5.3: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,q>1$, then

$$\int (e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,dx\,\longrightarrow$$

$$\frac{d\,(e\,x)^{m+1}\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^{q-1}}{b\,e\,(m+n\,(p+q)+1)}+\frac{1}{b\,(m+n\,(p+q)+1)}\int (e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^{q-2}\cdot$$
$$\left(c\,\left(\left(c\,b-a\,d\right)\,(m+1)+c\,b\,n\,(p+q)\right)+\left(d\,\left(c\,b-a\,d\right)\,(m+1)+d\,n\,(q-1)\,\left(b\,c-a\,d\right)+c\,b\,d\,n\,(p+q)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  d*(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)/(b*e*(m+n*(p+q)+1)) +
  1/(b*(m+n*(p+q)+1))*Int[(e*x)^m*(a+b*x^n)^p*(c+d*x^n)^(q-2)*
    Simp[c*((c*b-a*d)*(m+1)+c*b*n*(p+q))+(d*(c*b-a*d)*(m+1)+d*n*(q-1)*(b*c-a*d)+c*b*d*n*(p+q))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,p},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && GtQ[q,1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

4: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ q>0\ \wedge\ m-n+1>0$

Derivation: Binomial product recurrence 2b with $A=0$, $B=1$ and $m=m-n$

Derivation: Binomial product recurrence 4a with $A=c$, $B=d$ and $q=q-1$

Rule 1.1.3.4.6.1.5.4: If $b\,c-a\,d\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ q>0\ \wedge\ m-n+1>0$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow$$
$$\frac{e^{n-1}\,(e\,x)^{m-n+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^q}{b\,(m+n\,(p+q)+1)}-$$
$$\frac{e^n}{b\,(m+n\,(p+q)+1)}\int (e\,x)^{m-n}\left(a+b\,x^n\right)^p\left(c+d\,x^n\right)^{q-1}\left(a\,c\,(m-n+1)+\left(a\,d\,(m-n+1)-n\,q\,(b\,c-a\,d)\right)x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  e^(n-1)*(e*x)^(m-n+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(b*(m+n*(p+q)+1)) -
  e^n/(b*(m+n*(p+q)+1))*
    Int[(e*x)^(m-n)*(a+b*x^n)^p*(c+d*x^n)^(q-1)*Simp[a*c*(m-n+1)+(a*d*(m-n+1)-n*q*(b*c-a*d))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && GtQ[q,0] && GtQ[m-n+1,0] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

6: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m-n+1>n$

Derivation: Binomial product recurrence 4a with $A=0$, $B=1$ and $m=m-n$

Rule 1.1.3.4.6.1.6: If $b\,c-a\,d\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m-n+1>n$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow$$
$$\frac{e^{2n-1}\,(e\,x)^{m-2n+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q+1}}{b\,d\,(m+n\,(p+q)+1)}-\frac{e^{2n}}{b\,d\,(m+n\,(p+q)+1)}\,\cdot$$

$$\int (e\,x)^{m-2\,n}\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,\left(a\,c\,(m-2\,n+1)+\left(a\,d\,(m+n\,(q-1)+1)+b\,c\,(m+n\,(p-1)+1)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  e^(2*n-1)*(e*x)^(m-2*n+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(b*d*(m+n*(p+q)+1)) -
  e^(2*n)/(b*d*(m+n*(p+q)+1))*
    Int[(e*x)^(m-2*n)*(a+b*x^n)^p*(c+d*x^n)^q*Simp[a*c*(m-2*n+1)+(a*d*(m+n*(q-1)+1)+b*c*(m+n*(p-1)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,p,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && GtQ[m-n+1,n] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

7: $\int (e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m<-1$

Derivation: Binomial product recurrence 4b with $A=1$ and $B=0$

Rule 1.1.3.4.6.1.7: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m<-1$, then

$$\int (e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,dx\,\longrightarrow$$

$$\frac{(e\,x)^{m+1}\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^{q+1}}{a\,c\,e\,(m+1)}-$$

$$\frac{1}{a\,c\,e^n\,(m+1)}\int (e\,x)^{m+n}\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,\left(\left(b\,c+a\,d\right)\,(m+n+1)+n\,\left(b\,c\,p+a\,d\,q\right)+b\,d\,(m+n\,(p+q+2)+1)\,x^n\right)\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  (e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*c*e*(m+1)) -
  1/(a*c*e^n*(m+1))*
    Int[(e*x)^(m+n)*(a+b*x^n)^p*(c+d*x^n)^q*Simp[(b*c+a*d)*(m+n+1)+n*(b*c*p+a*d*q)+b*d*(m+n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,p,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LtQ[m,-1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

8. $\int \frac{(e\,x)^m\,(c+d\,x^n)^q}{a+b\,x^n}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+$

1. $\int \frac{(e\,x)^m}{(a+b\,x^n)\,(c+d\,x^n)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+$

1: $\int \frac{(e\,x)^m}{(a+b\,x^n)\,(c+d\,x^n)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,n\leq m\leq 2\,n-1$

Derivation: Algebraic expansion

Basis: If $n\in\mathbb{Z}$, then $\frac{(e\,x)^m}{(a+b\,x^n)\,(c+d\,x^n)} == -\frac{a\,e^n\,(e\,x)^{m-n}}{(b\,c-a\,d)\,(a+b\,x^n)}+\frac{c\,e^n\,(e\,x)^{m-n}}{(b\,c-a\,d)\,(c+d\,x^n)}$

Rule 1.1.3.4.6.1.8.1.1: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,n\leq m\leq 2\,n-1$, then

$$\int \frac{(e\,x)^m}{(a+b\,x^n)\,(c+d\,x^n)}\,dx \longrightarrow -\frac{a\,e^n}{b\,c-a\,d}\int \frac{(e\,x)^{m-n}}{a+b\,x^n}\,dx+\frac{c\,e^n}{b\,c-a\,d}\int \frac{(e\,x)^{m-n}}{c+d\,x^n}\,dx$$

Program code:

```
Int[(e_.*x_)^m_./((a_+b_.*x_^n_)*(c_+d_.*x_^n_)),x_Symbol] :=
  -a*e^n/(b*c-a*d)*Int[(e*x)^(m-n)/(a+b*x^n),x] + c*e^n/(b*c-a*d)*Int[(e*x)^(m-n)/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LeQ[n,m,2*n-1]
```

2: $\int \frac{(e\,x)^m}{(a+b\,x^n)\,(c+d\,x^n)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b\,z)\,(c+d\,z)}==\frac{b}{(b\,c-a\,d)\,(a+b\,z)}-\frac{d}{(b\,c-a\,d)\,(c+d\,z)}$

Rule 1.1.3.4.6.1.8.1.2: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+$, then

$$\int \frac{(e\,x)^m}{(a+b\,x^n)\,(c+d\,x^n)}\,dx\;\longrightarrow\;\frac{b}{b\,c-a\,d}\int \frac{(e\,x)^m}{a+b\,x^n}\,dx-\frac{d}{b\,c-a\,d}\int \frac{(e\,x)^m}{c+d\,x^n}\,dx$$

Program code:

```
Int[(e_.*x_)^m_./((a_+b_.*x_^n_)*(c_+d_.*x_^n_)),x_Symbol] :=
  b/(b*c-a*d)*Int[(e*x)^m/(a+b*x^n),x] - d/(b*c-a*d)*Int[(e*x)^m/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b*c-a*d,0] && IGtQ[n,0]
```

2: $\int \frac{(e\,x)^m\,(c+d\,x^n)^q}{a+b\,x^n}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,n\leq m\leq 2\,n-1$

Derivation: Algebraic expansion

Basis: If $n\in\mathbb{Z}$, then $\frac{1}{a+b\,x^n}==\frac{e^n\,(e\,x)^{-n}}{b}-\frac{a\,e^n\,(e\,x)^{-n}}{b\,(a+b\,x^n)}$

Rule 1.1.3.4.6.1.8.2: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,n\leq m\leq 2\,n-1$, then

$$\int \frac{(e\,x)^m\,(c+d\,x^n)^q}{a+b\,x^n}\,dx\,\longrightarrow\,\frac{e^n}{b}\int (e\,x)^{m-n}\,(c+d\,x^n)^q\,dx-\frac{a\,e^n}{b}\int \frac{(e\,x)^{m-n}\,(c+d\,x^n)^q}{a+b\,x^n}\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(c_+d_.*x_^n_)^q_./(a_+b_.*x_^n_),x_Symbol] :=
  e^n/b*Int[(e*x)^(m-n)*(c+d*x^n)^q,x] - a*e^n/b*Int[(e*x)^(m-n)*(c+d*x^n)^q/(a+b*x^n),x] /;
FreeQ[{a,b,c,d,e,m,q},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && LeQ[n,m,2*n-1] && IntBinomialQ[a,b,c,d,e,m,n,-1,q,x]
```

3. $\int \frac{x\left(a+b x^3\right)^q}{c+d x^3} \, dx$ when $b c-a d \neq 0 \wedge q^2 == \frac{1}{4} \wedge \left(b c-4 a d == 0 \vee b c+8 a d == 0 \vee b^2 c^2-20 a b c d-8 a^2 d^2 == 0\right)$

1. $\int \frac{x}{\left(c+d x^3\right) \sqrt{a+b x^3}} \, dx$ when $b c-a d \neq 0 \wedge \left(b c-4 a d == 0 \vee b c+8 a d == 0 \vee b^2 c^2-20 a b c d-8 a^2 d^2 == 0\right)$

1. $\int \frac{x}{\left(a+b x^3\right) \sqrt{c+d x^3}} \, dx$ when $b c-a d \neq 0 \wedge 4 b c-a d == 0$

1: $\int \frac{x}{\left(a+b x^3\right) \sqrt{c+d x^3}} \, dx$ when $b c-a d \neq 0 \wedge 4 b c-a d == 0 \wedge c > 0$

Reference: Goursat pseudo-elliptic integral

Contributed by Martin Welz on 24 January 2018 via sci.math.symbolic

Derivation: Algebraic expansion

Basis: If $4 b c-a d == 0 \wedge c > 0$, let $q \to \left(\frac{d}{c}\right)^{1/3}$, then

$$\frac{x}{\left(a+b x^3\right) \sqrt{c+d x^3}} == -\frac{q}{6 \times 2^{2/3} b x \sqrt{c+d x^3}}+\frac{d q x^2}{2^{5/3} b \left(4 c+d x^3\right) \sqrt{c+d x^3}}-\frac{q^2 \left(2^{2/3}-2 q x\right)}{12 b \left(2+2^{1/3} q x\right) \sqrt{c+d x^3}}+\frac{q \left(2^{4/3}+3 q^2 x^2-2^{1/3} q^3 x^3\right)}{6 \times 2^{2/3} b x \left(2^{4/3}-2^{2/3} q x+q^2 x^2\right) \sqrt{c+d x^3}}$$

Rule 1.1.3.4.6.1.8.3.1.1.1: If $b c-a d \neq 0 \wedge 4 b c-a d == 0 \wedge c > 0$, let $q \to \left(\frac{d}{c}\right)^{1/3}$, then

$$\int \frac{x}{\left(a+b x^3\right) \sqrt{c+d x^3}} \, dx \longrightarrow$$

$$-\int \frac{q}{6 \times 2^{2/3} b x \sqrt{c+d x^3}} \, dx+\int \frac{d q x^2}{2^{5/3} b \left(4 c+d x^3\right) \sqrt{c+d x^3}} \, dx-\int \frac{q^2 \left(2^{2/3}-2 q x\right)}{12 b \left(2+2^{1/3} q x\right) \sqrt{c+d x^3}} \, dx+\int \frac{q \left(2^{4/3}+3 q^2 x^2-2^{1/3} q^3 x^3\right)}{6 \times 2^{2/3} b x \left(2^{4/3}-2^{2/3} q x+q^2 x^2\right) \sqrt{c+d x^3}} \, dx \longrightarrow$$

$$\frac{q \operatorname{ArcTanh}\left[\frac{\sqrt{c+d x^3}}{\sqrt{c}}\right]}{9 \times 2^{2/3} b \sqrt{c}}+\frac{q \operatorname{ArcTan}\left[\frac{\sqrt{c+d x^3}}{\sqrt{3} \sqrt{c}}\right]}{3 \times 2^{2/3} \sqrt{3} b \sqrt{c}}-\frac{q \operatorname{ArcTan}\left[\frac{\sqrt{3} \sqrt{c} \left(1+2^{1/3} q x\right)}{\sqrt{c+d x^3}}\right]}{3 \times 2^{2/3} \sqrt{3} b \sqrt{c}}-\frac{q \operatorname{ArcTanh}\left[\frac{\sqrt{c} \left(1-2^{1/3} q x\right)}{\sqrt{c+d x^3}}\right]}{3 \times 2^{2/3} b \sqrt{c}}$$

Program code:

```
Int[x_/((a_+b_.*x_^3)*Sqrt[c_+d_.*x_^3]),x_Symbol] :=
  With[{q=Rt[d/c,3]},
  q*ArcTanh[Sqrt[c+d*x^3]/Rt[c,2]]/(9*2^(2/3)*b*Rt[c,2]) +
  q*ArcTan[Sqrt[c+d*x^3]/(Sqrt[3]*Rt[c,2])]/(3*2^(2/3)*Sqrt[3]*b*Rt[c,2]) -
  q*ArcTan[Sqrt[3]*Rt[c,2]*(1+2^(1/3)*q*x)/Sqrt[c+d*x^3]]/(3*2^(2/3)*Sqrt[3]*b*Rt[c,2]) -
  q*ArcTanh[Rt[c,2]*(1-2^(1/3)*q*x)/Sqrt[c+d*x^3]]/(3*2^(2/3)*b*Rt[c,2])] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[4*b*c-a*d,0] && PosQ[c]
```

2: $\int \frac{x}{(a+b x^3)\sqrt{c+d x^3}}\,dx$ when $b c-a d\neq 0 \wedge 4 b c-a d==0 \wedge c\ngtr 0$

Reference: Goursat pseudo-elliptic integral

Derivation: Algebraic expansion

Basis: If $4 b c-a d==0 \wedge c>0$, let $q\to\left(\frac{d}{c}\right)^{1/3}$, then

$$\frac{x}{(a+b x^3)\sqrt{c+d x^3}}==-\frac{q}{6\times 2^{2/3} b x\sqrt{c+d x^3}}+\frac{d q x^2}{2^{5/3} b (4 c+d x^3)\sqrt{c+d x^3}}-\frac{q^2 (2^{2/3}-2 q x)}{12 b (2+2^{1/3} q x)\sqrt{c+d x^3}}+\frac{q (2^{4/3}+3 q^2 x^2-2^{1/3} q^3 x^3)}{6\times 2^{2/3} b x (2^{4/3}-2^{2/3} q x+q^2 x^2)\sqrt{c+d x^3}}$$

Rule 1.1.3.4.6.1.8.3.1.1.2: If $b c-a d\neq 0 \wedge 4 b c-a d==0 \wedge c\ngtr 0$, let $q\to\left(\frac{d}{c}\right)^{1/3}$, then

$$\int \frac{x}{(a+b x^3)\sqrt{c+d x^3}}\,dx \longrightarrow$$

$$-\int \frac{q}{6\times 2^{2/3} b x\sqrt{c+d x^3}}\,dx+\int \frac{d q x^2}{2^{5/3} b (4 c+d x^3)\sqrt{c+d x^3}}\,dx-\int \frac{q^2 (2^{2/3}-2 q x)}{12 b (2+2^{1/3} q x)\sqrt{c+d x^3}}\,dx+\int \frac{q (2^{4/3}+3 q^2 x^2-2^{1/3} q^3 x^3)}{6\times 2^{2/3} b x (2^{4/3}-2^{2/3} q x+q^2 x^2)\sqrt{c+d x^3}}\,dx$$

$$\longrightarrow -\frac{q\,\mathrm{ArcTan}\left[\frac{\sqrt{c+d x^3}}{\sqrt{-c}}\right]}{9\times 2^{2/3} b\sqrt{-c}}-\frac{q\,\mathrm{ArcTanh}\left[\frac{\sqrt{c+d x^3}}{\sqrt{3}\sqrt{-c}}\right]}{3\times 2^{2/3}\sqrt{3} b\sqrt{-c}}-\frac{q\,\mathrm{ArcTanh}\left[\frac{\sqrt{3}\sqrt{-c}(1+2^{1/3} q x)}{\sqrt{c+d x^3}}\right]}{3\times 2^{2/3}\sqrt{3} b\sqrt{-c}}-\frac{q\,\mathrm{ArcTan}\left[\frac{\sqrt{-c}(1-2^{1/3} q x)}{\sqrt{c+d x^3}}\right]}{3\times 2^{2/3} b\sqrt{-c}}$$

Program code:

```
Int[x_/((a_+b_.*x_^3)*Sqrt[c_+d_.*x_^3]),x_Symbol] :=
  With[{q=Rt[d/c,3]},
  -q*ArcTan[Sqrt[c+d*x^3]/Rt[-c,2]]/(9*2^(2/3)*b*Rt[-c,2]) -
  q*ArcTanh[Sqrt[c+d*x^3]/(Sqrt[3]*Rt[-c,2])]/(3*2^(2/3)*Sqrt[3]*b*Rt[-c,2]) -
  q*ArcTanh[Sqrt[3]*Rt[-c,2]*(1+2^(1/3)*q*x)/Sqrt[c+d*x^3]]/(3*2^(2/3)*Sqrt[3]*b*Rt[-c,2]) -
  q*ArcTan[Rt[-c,2]*(1-2^(1/3)*q*x)/Sqrt[c+d*x^3]]/(3*2^(2/3)*b*Rt[-c,2])] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[4*b*c-a*d,0] && NegQ[c]
```

2: $\int \frac{x}{(a+b x^3) \sqrt{c+d x^3}} \, dx$ when $b c - a d \neq 0 \land 8 b c + a d == 0$

Reference: Goursat pseudo-elliptic integral

Contributed by Martin Welz on 22 January 2018 via sci.math.symbolic

Derivation: Algebraic expansion

Basis: If $8 b c + a d == 0$, let $q \to \left(\frac{d}{c}\right)^{1/3}$, then $\frac{x}{a+b x^3} == \frac{d q x^2}{4 b (8 c-d x^3)} - \frac{q^2 (1+q x)}{12 b (2-q x)} + \frac{2 c q^2-2 d x-d q x^2}{12 b c (4+2 q x+q^2 x^2)}$

Rule 1.1.3.4.6.1.8.3.1.2: If $b c - a d \neq 0 \land 8 b c + a d == 0$, let $q \to \left(\frac{d}{c}\right)^{1/3}$, then

$$\int \frac{x}{(a+b x^3) \sqrt{c+d x^3}} \, dx \longrightarrow \frac{d q}{4 b} \int \frac{x^2}{(8 c-d x^3) \sqrt{c+d x^3}} \, dx - \frac{q^2}{12 b} \int \frac{1+q x}{(2-q x) \sqrt{c+d x^3}} \, dx + \frac{1}{12 b c} \int \frac{2 c q^2-2 d x-d q x^2}{(4+2 q x+q^2 x^2) \sqrt{c+d x^3}} \, dx$$

Program code:

```
Int[x_/((a_+b_.*x_^3)*Sqrt[c_+d_.*x_^3]),x_Symbol] :=
  With[{q=Rt[d/c,3]},
  d*q/(4*b)*Int[x^2/((8*c-d*x^3)*Sqrt[c+d*x^3]),x] -
  q^2/(12*b)*Int[(1+q*x)/((2-q*x)*Sqrt[c+d*x^3]),x] +
  1/(12*b*c)*Int[(2*c*q^2-2*d*x-d*q*x^2)/((4+2*q*x+q^2*x^2)*Sqrt[c+d*x^3]),x]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[8*b*c+a*d,0]
```

3. $\int \frac{x}{(c+d x^3) \sqrt{a+b x^3}} \, dx$ when $b c - a d \neq 0 \land b^2 c^2 - 20 a b c d - 8 a^2 d^2 == 0$

1: $\int \frac{x}{(c+d x^3) \sqrt{a+b x^3}} \, dx$ when $b c - a d \neq 0 \land b^2 c^2 - 20 a b c d - 8 a^2 d^2 == 0 \land a > 0$

Reference: Goursat pseudo-elliptic integral

Note: If $b^2 c^2 - 20 a b c d - 8 a^2 d^2 == \left(b c - 10 a d + 6 \sqrt{3} a d\right) \left(b c - 10 a d - 6 \sqrt{3} a d\right) == 0$, then $\frac{b c-10 a d}{6 a d}$ should simplify to $\sqrt{3}$ or $-\sqrt{3}$.

Rule 1.1.3.4.6.1.8.3.1.3.1: If $b c - a d \neq 0 \land b^2 c^2 - 20 a b c d - 8 a^2 d^2 == 0 \land a > 0$, let $q \to \left(\frac{b}{a}\right)^{1/3}$ and $r \to \frac{b c-10 a d}{6 a d}$, then

$$\int \frac{x}{\left(c + d x^3\right) \sqrt{a + b x^3}} \, dx \longrightarrow$$

$$-\frac{q (2-r) \operatorname{ArcTan}\left[\frac{(1-r) \sqrt{a+b x^3}}{\sqrt{2} \sqrt{a} r^{3/2}}\right]}{3 \sqrt{2} \sqrt{a} d r^{3/2}} - \frac{q (2-r) \operatorname{ArcTan}\left[\frac{\sqrt{a} \sqrt{r} (1+r) (1+q x)}{\sqrt{2} \sqrt{a+b x^3}}\right]}{2 \sqrt{2} \sqrt{a} d r^{3/2}} - \frac{q (2-r) \operatorname{ArcTanh}\left[\frac{\sqrt{a} (1-r) \sqrt{r} (1+q x)}{\sqrt{2} \sqrt{a+b x^3}}\right]}{6 \sqrt{2} \sqrt{a} d \sqrt{r}} - \frac{q (2-r) \operatorname{ArcTanh}\left[\frac{\sqrt{a} \sqrt{r} (1+r-2 q x)}{\sqrt{2} \sqrt{a+b x^3}}\right]}{3 \sqrt{2} \sqrt{a} d \sqrt{r}}$$

Program code:

```
Int[x_/((c_+d_.*x_^3)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{q=Rt[b/a,3],r=Simplify[(b*c-10*a*d)/(6*a*d)]},
  -q*(2-r)*ArcTan[(1-r)*Sqrt[a+b*x^3]/(Sqrt[2]*Rt[a,2]*r^(3/2))]/(3*Sqrt[2]*Rt[a,2]*d*r^(3/2)) -
  q*(2-r)*ArcTan[Rt[a,2]*Sqrt[r]*(1+r)*(1+q*x)/(Sqrt[2]*Sqrt[a+b*x^3])]/(2*Sqrt[2]*Rt[a,2]*d*r^(3/2)) -
  q*(2-r)*ArcTanh[Rt[a,2]*(1-r)*Sqrt[r]*(1+q*x)/(Sqrt[2]*Sqrt[a+b*x^3])]/(6*Sqrt[2]*Rt[a,2]*d*Sqrt[r]) -
  q*(2-r)*ArcTanh[Rt[a,2]*Sqrt[r]*(1+r-2*q*x)/(Sqrt[2]*Sqrt[a+b*x^3])]/(3*Sqrt[2]*Rt[a,2]*d*Sqrt[r])] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b^2*c^2-20*a*b*c*d-8*a^2*d^2,0] && PosQ[a]
```

2: $\int \frac{x}{\left(c + d x^3\right) \sqrt{a + b x^3}} \, dx$ when $b c - a d \neq 0 \land b^2 c^2 - 20 a b c d - 8 a^2 d^2 == 0 \land a \ngtr 0$

Reference: Goursat pseudo-elliptic integral

Note: If $b^2 c^2 - 20 a b c d - 8 a^2 d^2 == \left(b c - 10 a d + 6 \sqrt{3} a d\right) \left(b c - 10 a d - 6 \sqrt{3} a d\right) == 0$, then $\frac{b c-10 a d}{6 a d}$ should simplify to $\sqrt{3}$ or $-\sqrt{3}$.

Rule 1.1.3.4.6.1.8.3.1.3.2: If $b c - a d \neq 0 \land b^2 c^2 - 20 a b c d - 8 a^2 d^2 == 0 \land a \ngtr 0$, let $q \to \left(\frac{b}{a}\right)^{1/3}$, and $r \to \frac{b c-10 a d}{6 a d}$, then

$$\int \frac{x}{\left(c+d\,x^3\right)\sqrt{a+b\,x^3}}\,dx \longrightarrow$$

$$\frac{q\,(2-r)\,\text{ArcTanh}\left[\frac{(1-r)\sqrt{a+b\,x^3}}{\sqrt{2}\,\sqrt{-a}\,r^{3/2}}\right]}{3\sqrt{2}\,\sqrt{-a}\,d\,r^{3/2}}-\frac{q\,(2-r)\,\text{ArcTanh}\left[\frac{\sqrt{-a}\,\sqrt{r}\,(1+r)\,(1+q\,x)}{\sqrt{2}\,\sqrt{a+b\,x^3}}\right]}{2\sqrt{2}\,\sqrt{-a}\,d\,r^{3/2}}-\frac{q\,(2-r)\,\text{ArcTan}\left[\frac{\sqrt{-a}\,(1-r)\sqrt{r}\,(1+q\,x)}{\sqrt{2}\,\sqrt{a+b\,x^3}}\right]}{6\sqrt{2}\,\sqrt{-a}\,d\,\sqrt{r}}-\frac{q\,(2-r)\,\text{ArcTan}\left[\frac{\sqrt{-a}\,\sqrt{r}\,(1+r-2\,q\,x)}{\sqrt{2}\,\sqrt{a+b\,x^3}}\right]}{3\sqrt{2}\,\sqrt{-a}\,d\,\sqrt{r}}$$

Program code:

```
Int[x_/((c_+d_.*x_^3)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{q=Rt[b/a,3],r=Simplify[(b*c-10*a*d)/(6*a*d)]},
  q*(2-r)*ArcTanh[(1-r)*Sqrt[a+b*x^3]/(Sqrt[2]*Rt[-a,2]*r^(3/2))]/(3*Sqrt[2]*Rt[-a,2]*d*r^(3/2)) -
  q*(2-r)*ArcTanh[Rt[-a,2]*Sqrt[r]*(1+r)*(1+q*x)/(Sqrt[2]*Sqrt[a+b*x^3])]/(2*Sqrt[2]*Rt[-a,2]*d*r^(3/2)) -
  q*(2-r)*ArcTan[Rt[-a,2]*(1-r)*Sqrt[r]*(1+q*x)/(Sqrt[2]*Sqrt[a+b*x^3])]/(6*Sqrt[2]*Rt[-a,2]*d*Sqrt[r]) -
  q*(2-r)*ArcTan[Rt[-a,2]*Sqrt[r]*(1+r-2*q*x)/(Sqrt[2]*Sqrt[a+b*x^3])]/(3*Sqrt[2]*Rt[-a,2]*d*Sqrt[r])] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b^2*c^2-20*a*b*c*d-8*a^2*d^2,0] && NegQ[a]
```

2: $\int \frac{x\sqrt{a+b x^3}}{c+d x^3}\,dx$ when $b c-a d\neq 0 \land (b c-4 a d=0 \lor b c+8 a d=0 \lor b^2 c^2-20 a b c d-8 a^2 d^2=0)$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{c+d z}=\frac{b}{d\sqrt{a+b z}}-\frac{b c-a d}{d (c+d z)\sqrt{a+b z}}$

Rule 1.1.3.4.6.1.8.3.2: If $b c-a d\neq 0 \land (b c-4 a d=0 \lor b c+8 a d=0 \lor b^2 c^2-20 a b c d-8 a^2 d^2=0)$, then

$$\int \frac{x\sqrt{a+b x^3}}{c+d x^3}\,dx \rightarrow \frac{b}{d}\int \frac{x}{\sqrt{a+b x^3}}\,dx-\frac{b c-a d}{d}\int \frac{x}{(c+d x^3)\sqrt{a+b x^3}}\,dx$$

Program code:

```
Int[x_*Sqrt[a_+b_.*x_^3]/(c_+d_.*x_^3),x_Symbol] :=
  b/d*Int[x/Sqrt[a+b*x^3],x] - (b*c-a*d)/d*Int[x/((c+d*x^3)*Sqrt[a+b*x^3]),x] /;
FreeQ[{c,d,a,b},x] && NeQ[b*c-a*d,0] && (EqQ[b*c-4*a*d,0] || EqQ[b*c+8*a*d,0] || EqQ[b^2*c^2-20*a*b*c*d-8*a^2*d^2,0])
```

4. $\int \frac{x^2 (c+d x^4)^q}{a+b x^4}\,dx$ when $b c-a d\neq 0 \land q^2=\frac{1}{4}$

1: $\int \frac{x^2}{(a+b x^4)\sqrt{c+d x^4}}\,dx$ when $b c-a d\neq 0$

Derivation: Algebraic expansion

Basis: If $\frac{r}{s}=\left(-\frac{a}{b}\right)^{1/2}$, then $\frac{x^2}{a+b x^4}=\frac{s}{2 b (r+s x^2)}-\frac{s}{2 b (r-s x^2)}$

Rule 1.1.3.4.6.1.8.4.1: If $b c-a d\neq 0$, let $\frac{r}{s}=\left(-\frac{a}{b}\right)^{1/2}$, then

$$\int \frac{x^2}{\left(a+b\,x^4\right)\sqrt{c+d\,x^4}}\,dx \to \frac{s}{2\,b}\int \frac{1}{\left(r+s\,x^2\right)\sqrt{c+d\,x^4}}\,dx-\frac{s}{2\,b}\int \frac{1}{\left(r-s\,x^2\right)\sqrt{c+d\,x^4}}\,dx$$

Program code:

```
Int[x_^2/((a_+b_.*x_^4)*Sqrt[c_+d_.*x_^4]),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  s/(2*b)*Int[1/((r+s*x^2)*Sqrt[c+d*x^4]),x] - s/(2*b)*Int[1/((r-s*x^2)*Sqrt[c+d*x^4]),x]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

2: $\int \frac{x^2\sqrt{c+d\,x^4}}{a+b\,x^4}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{c+d\,z}}{a+b\,z}=\frac{d}{b\,\sqrt{c+d\,z}}+\frac{b\,c-a\,d}{b\,(a+b\,z)\,\sqrt{c+d\,z}}$

Rule 1.1.3.4.6.1.8.4.2: If $b\,c-a\,d\neq 0$, then

$$\int \frac{x^2\sqrt{c+d\,x^4}}{a+b\,x^4}\,dx \to \frac{d}{b}\int \frac{x^2}{\sqrt{c+d\,x^4}}\,dx+\frac{b\,c-a\,d}{b}\int \frac{x^2}{\left(a+b\,x^4\right)\sqrt{c+d\,x^4}}\,dx$$

Program code:

```
Int[x_^2*Sqrt[c_+d_.*x_^4]/(a_+b_.*x_^4),x_Symbol] :=
  d/b*Int[x^2/Sqrt[c+d*x^4],x] + (b*c-a*d)/b*Int[x^2/((a+b*x^4)*Sqrt[c+d*x^4]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

9. $\int \frac{x^m}{\sqrt{a+b\,x^n}\,\sqrt{c+d\,x^n}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,(m\mid n)\in\mathbb{Z}\,\land\,0<m-n+1<n$

1: $\int \frac{x^2}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}\, dx$ when $b c-a d \neq 0 \land \frac{b}{a}>0 \land \frac{d}{c}>0$

Rule 1.1.3.4.6.1.9.1: If $b c-a d \neq 0 \land \frac{b}{a}>0 \land \frac{d}{c}>0$, then

$$\int \frac{x^2}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}\, dx \longrightarrow \frac{x\sqrt{a+b x^2}}{b\sqrt{c+d x^2}}-\frac{c}{b}\int \frac{\sqrt{a+b x^2}}{\left(c+d x^2\right)^{3/2}}\, dx$$

Program code:

```
Int[x_^2/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  x*Sqrt[a+b*x^2]/(b*Sqrt[c+d*x^2]) - c/b*Int[Sqrt[a+b*x^2]/(c+d*x^2)^(3/2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && PosQ[b/a] && PosQ[d/c] && Not[SimplerSqrtQ[b/a,d/c]]
```

2: $\int \frac{x^n}{\sqrt{a+b x^n}\ \sqrt{c+d x^n}}\, dx$ when $b c-a d \neq 0 \land (n == 2 \lor n == 4)$

Derivation: Algebraic expansion

Basis: $\frac{z}{\sqrt{a+b z}} == \frac{\sqrt{a+b z}}{b}-\frac{a}{b\sqrt{a+b z}}$

Rule 1.1.3.4.6.1.9.2: If $b c-a d \neq 0 \land (n == 2 \lor n == 4)$, then

$$\int \frac{x^n}{\sqrt{a+b x^n}\ \sqrt{c+d x^n}}\, dx \longrightarrow \frac{1}{b}\int \frac{\sqrt{a+b x^n}}{\sqrt{c+d x^n}}\, dx-\frac{a}{b}\int \frac{1}{\sqrt{a+b x^n}\ \sqrt{c+d x^n}}\, dx$$

Program code:

```
Int[x_^n_/(Sqrt[a_+b_.*x_^n_]*Sqrt[c_+d_.*x_^n_]),x_Symbol] :=
  1/b*Int[Sqrt[a+b*x^n]/Sqrt[c+d*x^n],x] - a/b*Int[1/(Sqrt[a+b*x^n]*Sqrt[c+d*x^n]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && (EqQ[n,2] || EqQ[n,4]) && Not[EqQ[n,2] && SimplerSqrtQ[-b/a,-d/c]]
```

10: $\int x^m (a+b x^n)^p (c+d x^n)^q \, dx$ when $n \in \mathbb{Z}^+ \land \left(p+\frac{m+1}{n} \mid q\right) \in \mathbb{Z} \land -1 < p < 0$

Derivation: Integration by substitution

Basis: If $p+\frac{m+1}{n} \in \mathbb{Z}$, let $k = \text{Denominator}[p]$, then

$$x^m (a+b x^n)^p F[x^n] = \frac{k\, a^{p+\frac{m+1}{n}}}{n} \text{Subst}\left[\frac{x^{\frac{k(m+1)}{n}-1}}{(1-b x^k)^{p+\frac{m+1}{n}+1}} F\left[\frac{a x^k}{1-b x^k}\right], x, \frac{x^{n/k}}{(a+b x^n)^{1/k}}\right] \partial_x \frac{x^{n/k}}{(a+b x^n)^{1/k}}$$

Basis: If $\left(p+\frac{m+1}{n} \mid q\right) \in \mathbb{Z}$, let $k = \text{Denominator}[p]$, then

$$x^m (a+b x^n)^p (c+d x^n)^q = \frac{k\, a^{p+\frac{m+1}{n}}}{n} \text{Subst}\left[\frac{x^{\frac{k(m+1)}{n}-1} (c-(b c-a d) x^k)^q}{(1-b x^k)^{p+q+\frac{m+1}{n}+1}}, x, \frac{x^{n/k}}{(a+b x^n)^{1/k}}\right] \partial_x \frac{x^{n/k}}{(a+b x^n)^{1/k}}$$

Note: The exponents in the resulting integrand are integers.

Rule 1.1.3.4.6.1.10: If $n \in \mathbb{Z}^+ \land \left(p+\frac{m+1}{n} \mid q\right) \in \mathbb{Z} \land -1 < p < 0$, let $k = \text{Denominator}[p]$, then

$$\int x^m (a+b x^n)^p (c+d x^n)^q \, dx \to \frac{k\, a^{p+\frac{m+1}{n}}}{n} \text{Subst}\left[\int \frac{x^{\frac{k(m+1)}{n}-1} \left(c-(b c-a d) x^k\right)^q}{(1-b x^k)^{p+q+\frac{m+1}{n}+1}} dx, x, \frac{x^{n/k}}{(a+b x^n)^{1/k}}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  With[{k=Denominator[p]},
  k*a^(p+(m+1)/n)/n*
    Subst[Int[x^(k*(m+1)/n-1)*(c-(b*c-a*d)*x^k)^q/(1-b*x^k)^(p+q+(m+1)/n+1),x],x,x^(n/k)/(a+b*x^n)^(1/k)]] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,0] && RationalQ[m,p] && IntegersQ[p+(m+1)/n,q] && LtQ[-1,p,0]
```

2. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\wedge n\in\mathbb{Z}^-$

1. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\wedge n\in\mathbb{Z}^-\wedge m\in\mathbb{Q}$

1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\wedge n\in\mathbb{Z}^-\wedge m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $F[x] = -\mathrm{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule 1.1.3.4.6.2.1.1: If $b\,c-a\,d\neq 0\wedge n\in\mathbb{Z}^-\wedge m\in\mathbb{Z}$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\longrightarrow\,-\mathrm{Subst}\left[\int\frac{(a+b\,x^{-n})^p\,(c+d\,x^{-n})^q}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n))^p*(c+d*x^(-n))^q/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,p,q},x] && NeQ[b*c-a*d,0] && ILtQ[n,0] && IntegerQ[m]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \, dx$ when $b\,c-a\,d\neq 0 \land n\in\mathbb{Z}^- \land m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z} \land g>1$, then $(e\,x)^m F[x^n] = -\frac{g}{e}\,\text{Subst}\left[\frac{F\left[e^{-n}\,x^{-g\,n}\right]}{x^{g\,(m+1)+1}},\,x,\,\frac{1}{(e\,x)^{1/g}}\right]\partial_x \frac{1}{(e\,x)^{1/g}}$

Rule 1.1.3.4.6.2.1.2: If $b\,c-a\,d\neq 0 \land n\in\mathbb{Z}^- \land m\in\mathbb{F}$, let $g=\text{Denominator}[m]$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx \to -\frac{g}{e}\,\text{Subst}\left[\int \frac{\left(a+b\,e^{-n}\,x^{-g\,n}\right)^p \left(c+d\,e^{-n}\,x^{-g\,n}\right)^q}{x^{g\,(m+1)+1}}\,dx,\,x,\,\frac{1}{(e\,x)^{1/g}}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  With[{g=Denominator[m]},
  -g/e*Subst[Int[(a+b*e^(-n)*x^(-g*n))^p*(c+d*e^(-n)*x^(-g*n))^q/x^(g*(m+1)+1),x],x,1/(e*x)^(1/g)]] /;
FreeQ[{a,b,c,d,e,p,q},x] && ILtQ[n,0] && FractionQ[m]
```

2: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^-\wedge\,m\notin\mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\left((e\,x)^m\,\left(x^{-1}\right)^m\right)=0$

Basis: $F[x]=-\text{Subst}\left[\frac{F\left[x^{-1}\right]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule 1.1.3.4.6.2.2: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^-\wedge\,m\notin\mathbb{Q}$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\,\rightarrow\,(e\,x)^m\,\left(x^{-1}\right)^m\int\frac{(a+b\,x^n)^p\,(c+d\,x^n)^q}{\left(x^{-1}\right)^m}\,dx$$

$$\rightarrow\,-(e\,x)^m\,\left(x^{-1}\right)^m\,\text{Subst}\left[\int\frac{\left(a+b\,x^{-n}\right)^p\,\left(c+d\,x^{-n}\right)^q}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -(e*x)^m*(x^(-1))^m*Subst[Int[(a+b*x^(-n))^p*(c+d*x^(-n))^q/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,m,p,q},x] && NeQ[b*c-a*d,0] && ILtQ[n,0] && Not[RationalQ[m]]
```

7. $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{F}$

1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g\in\mathbb{Z}^+$, then $x^m\,F[x^n] = g\,\text{Subst}\left[x^{g\,(m+1)-1}\,F[x^{g\,n}],\,x,\,x^{1/g}\right]\partial_x x^{1/g}$

Rule 1.1.3.4.7.1: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{F}$, let $g=\text{Denominator}[n]$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow\;g\,\text{Subst}\left[\int x^{g\,(m+1)-1}\,(a+b\,x^{g\,n})^p\,(c+d\,x^{g\,n})^q\,dx,\,x,\,x^{1/g}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g*(m+1)-1)*(a+b*x^(g*n))^p*(c+d*x^(g*n))^q,x],x,x^(1/g)]] /;
FreeQ[{a,b,c,d,m,p,q},x] && NeQ[b*c-a*d,0] && FractionQ[n]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\land n\in\mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m}==0$

Basis: $\frac{(e\,x)^m}{x^m}==\frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.4.7.2: If $b\,c-a\,d\neq 0\land n\in\mathbb{F}$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$$

Program code:

```
Int[(e_*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,m,p,q},x] && NeQ[b*c-a*d,0] && FractionQ[n]
```

8. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\ \land\ \frac{n}{m+1}\in\mathbb{Z}$

x: $\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\ \land\ \frac{n}{m+1}\in\mathbb{Z}^-\ \land\ m\neq -1\ \land\ -1\leq p<0\ \land\ -1\leq q<0$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1}\in\mathbb{Z}$, then $x^m\,F[x^n] == -\frac{1}{m+1}\,\frac{F\left[\left(x^{-(m+1)}\right)^{-\frac{n}{m+1}}\right]}{\left(x^{-(m+1)}\right)^2}\,\partial_x\,x^{-(m+1)}$

Rule 1.1.3.4.8.x: If $b\,c-a\,d\neq 0\ \land\ m\neq -1\ \land\ \frac{n}{m+1}\in\mathbb{Z}^-\ \land\ -1\leq p<0\ \land\ -1\leq q<0$, then

$$\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow -\frac{1}{m+1}\,\text{Subst}\left[\int \frac{\left(a+b\,x^{-\frac{n}{m+1}}\right)^p \left(c+d\,x^{-\frac{n}{m+1}}\right)^q}{x^2}\,dx,\ x,\ x^{-(m+1)}\right]$$

Program code:

```
(* Int[x_^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -1/(m+1)*Subst[Int[(a+b*x^Simplify[-n/(m+1)])^p*(c+d*x^Simplify[-n/(m+1)])^q/x^2,x],x,x^(-(m+1))] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && NeQ[m,-1] && ILtQ[Simplify[n/(m+1)+1],0] &&
  GeQ[p,-1] && LtQ[p,0] && GeQ[q,-1] && LtQ[q,0] && Not[IntegerQ[n]] *)
```

1: $\int x^m (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land \frac{n}{m+1} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1} \in \mathbb{Z}$, then $x^m F[x^n] = \frac{1}{m+1} \text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right], x, x^{m+1}\right] \partial_x x^{m+1}$

Rule 1.1.3.4.8.1: If $b c - a d \neq 0 \land \frac{n}{m+1} \in \mathbb{Z}$, then

$$\int x^m (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow \frac{1}{m+1} \text{Subst}\left[\int \left(a+b x^{\frac{n}{m+1}}\right)^p \left(c+d x^{\frac{n}{m+1}}\right)^q dx, x, x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*x^Simplify[n/(m+1)])^p*(c+d*x^Simplify[n/(m+1)])^q,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && NeQ[b*c-a*d,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0\ \wedge\ \frac{n}{m+1}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Basis: $\frac{(e\,x)^m}{x^m} == \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.4.8.2: If $b\,c-a\,d\neq 0\ \wedge\ \frac{n}{m+1}\in\mathbb{Z}$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && NeQ[b*c-a*d,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

9. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0\ \wedge\ p<-1$

1. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0\ \wedge\ p<-1\ \wedge\ q>0$

1: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0\ \wedge\ p<-1\ \wedge\ q>1$

Derivation: Binomial product recurrence 1 with $A=c$, $B=d$ and $q=q-1$

Rule 1.1.3.4.9.1.1: If $b\,c-a\,d\neq 0\ \wedge\ p<-1\ \wedge\ q>1$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx \longrightarrow$$
$$\frac{-\left(c\,b-a\,d\right)\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q-1}}{a\,b\,e\,n\,(p+1)}+\frac{1}{a\,b\,n\,(p+1)}\cdot$$

$$\int (e\,x)^m \left(a+b\,x^n\right)^{p+1} \left(c+d\,x^n\right)^{q-2} \left(c\,\left(c\,b\,n\,(p+1)+\left(c\,b-a\,d\right)\,(m+1)\right)+d\,\left(c\,b\,n\,(p+1)+\left(c\,b-a\,d\right)\,(m+n\,(q-1)+1)\right)\,x^n\right)\,dx$$

Programcode:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -(c*b-a*d)*(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)/(a*b*e*n*(p+1)) +
  1/(a*b*n*(p+1))*Int[(e*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-2)*
    Simp[c*(c*b*n*(p+1)+(c*b-a*d)*(m+1))+d*(c*b*n*(p+1)+(c*b-a*d)*(m+n*(q-1)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && GtQ[q,1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx$ when $b\,c-a\,d\neq 0\ \land\ p<-1\ \land\ 0<q<1$

Derivation: Binomial product recurrence 1 with $A = 1$ and $B = 0$

Derivation: Binomial product recurrence 3b with $A = c$, $B = d$ and $q = q-1$

Rule 1.1.3.4.9.1.2: If $b\,c-a\,d\neq 0\ \land\ p<-1\ \land\ 0<q<1$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q\,dx \longrightarrow$$
$$-\frac{(e\,x)^{m+1} \left(a+b\,x^n\right)^{p+1} \left(c+d\,x^n\right)^q}{a\,e\,n\,(p+1)}+$$
$$\frac{1}{a\,n\,(p+1)} \int (e\,x)^m \left(a+b\,x^n\right)^{p+1} \left(c+d\,x^n\right)^{q-1} \left(c\,(m+n\,(p+1)+1)+d\,(m+n\,(p+q+1)+1)\,x^n\right)\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*e*n*(p+1)) +
  1/(a*n*(p+1))*Int[(e*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*Simp[c*(m+n*(p+1)+1)+d*(m+n*(p+q+1)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && LtQ[0,q,1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx$ when $b\,c-a\,d\neq 0\,\wedge\,p<-1$

Derivation: Binomial product recurrence 3b with $A=1$ and $B=0$

Rule 1.1.3.4.9.2: If $b\,c-a\,d\neq 0\,\wedge\,p<-1$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx \longrightarrow$$
$$-\frac{b\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q+1}}{a\,e\,n\,\left(b\,c-a\,d\right)\,(p+1)}+$$
$$\frac{1}{a\,n\,\left(b\,c-a\,d\right)\,(p+1)}\int (e\,x)^m \left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^q \left(c\,b\,(m+1)+n\,\left(b\,c-a\,d\right)\,(p+1)+d\,b\,(m+n\,(p+q+2)+1)\,x^n\right) dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -b*(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*e*n*(b*c-a*d)*(p+1)) +
  1/(a*n*(b*c-a*d)*(p+1))*
    Int[(e*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^q*Simp[c*b*(m+1)+n*(b*c-a*d)*(p+1)+d*b*(m+n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,n,q},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

10. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx$ when $b\,c-a\,d\neq 0\,\wedge\,q>0$

1: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx$ when $b\,c-a\,d\neq 0\,\wedge\,q>0\,\wedge\,p>0$

Derivation: Binomial product recurrence 2b with $A=a$, $B=b$ and $p=p-1$

Rule 1.1.3.4.10.1: If $b\,c-a\,d\neq 0\,\wedge\,q>0\,\wedge\,p>0$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q dx \longrightarrow$$
$$\frac{(e\,x)^{m+1}\left(a+b\,x^n\right)^p\left(c+d\,x^n\right)^q}{e\,(m+n\,(p+q)+1)}+$$

$$\frac{n}{m+n(p+q)+1}\int(e\,x)^m\left(a+b\,x^n\right)^{p-1}\left(c+d\,x^n\right)^{q-1}\left(a\,c\,(p+q)+\left(q\left(b\,c-a\,d\right)+a\,d\,(p+q)\right)x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  (e*x)^(m+1)*(a+b*x^n)^p*(c+d*x^n)^q/(e*(m+n*(p+q)+1)) +
  n/(m+n*(p+q)+1)*Int[(e*x)^m*(a+b*x^n)^(p-1)*(c+d*x^n)^(q-1)*Simp[a*c*(p+q)+(q*(b*c-a*d)+a*d*(p+q))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b*c-a*d,0] && GtQ[q,0] && GtQ[p,0] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

2: $\int(e\,x)^m\left(a+b\,x^n\right)^p\left(c+d\,x^n\right)^q dx$ when $b\,c-a\,d\neq 0\wedge q>1$

Derivation: Binomial product recurrence 2b with A = c, B = d and q = q − 1

Rule 1.1.3.4.10.2: If $b\,c-a\,d\neq 0\wedge q>1$, then

$$\int(e\,x)^m\left(a+b\,x^n\right)^p\left(c+d\,x^n\right)^q dx\,\longrightarrow$$

$$\frac{d\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q-1}}{b\,e\,(m+n\,(p+q)+1)}+\frac{1}{b\,(m+n\,(p+q)+1)}\int(e\,x)^m\left(a+b\,x^n\right)^p\left(c+d\,x^n\right)^{q-2}\cdot$$
$$\left(c\left(\left(c\,b-a\,d\right)(m+1)+c\,b\,n\,(p+q)\right)+\left(d\left(c\,b-a\,d\right)(m+1)+d\,n\,(q-1)\left(b\,c-a\,d\right)+c\,b\,d\,n\,(p+q)\right)x^n\right)dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  d*(e*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)/(b*e*(m+n*(p+q)+1)) +
  1/(b*(m+n*(p+q)+1))*Int[(e*x)^m*(a+b*x^n)^p*(c+d*x^n)^(q-2)*
    Simp[c*((c*b-a*d)*(m+1)+c*b*n*(p+q))+(d*(c*b-a*d)*(m+1)+d*n*(q-1)*(b*c-a*d)+c*b*d*n*(p+q))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[b*c-a*d,0] && GtQ[q,1] && IntBinomialQ[a,b,c,d,e,m,n,p,q,x]
```

11. $\int\frac{(e\,x)^m}{\left(a+b\,x^n\right)\left(c+d\,x^n\right)}dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{x^m}{(a+b x^n)(c+d x^n)} dx$ when $b c - a d \neq 0 \wedge (m == n \vee m == 2 n - 1)$

Derivation: Algebraic expansion

Basis: $\frac{x^m}{(a+b x^n)(c+d x^n)} == -\frac{a x^{m-n}}{(b c-a d)(a+b x^n)} + \frac{c x^{m-n}}{(b c-a d)(c+d x^n)}$

Rule 1.1.3.4.11.1: If $b c - a d \neq 0 \wedge (m == n \vee m == 2 n - 1)$, then

$$\int \frac{x^m}{(a+b x^n)(c+d x^n)} dx \to -\frac{a}{b c - a d} \int \frac{x^{m-n}}{a+b x^n} dx + \frac{c}{b c - a d} \int \frac{x^{m-n}}{c+d x^n} dx$$

Program code:

```
Int[x_^m_/((a_+b_.*x_^n_)*(c_+d_.*x_^n_)),x_Symbol] :=
  -a/(b*c-a*d)*Int[x^(m-n)/(a+b*x^n),x] + c/(b*c-a*d)*Int[x^(m-n)/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && (EqQ[m,n] || EqQ[m,2*n-1])
```

2: $\int \frac{(e x)^m}{(a+b x^n)(c+d x^n)} dx$ when $b c - a d \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b z)(c+d z)} == \frac{b}{(b c-a d)(a+b z)} - \frac{d}{(b c-a d)(c+d z)}$

Rule 1.1.3.4.11.2: If $b c - a d \neq 0$, then

$$\int \frac{(e x)^m}{(a+b x^n)(c+d x^n)} dx \to \frac{b}{b c - a d} \int \frac{(e x)^m}{a+b x^n} dx - \frac{d}{b c - a d} \int \frac{(e x)^m}{c+d x^n} dx$$

Program code:

```
Int[(e_.*x_)^m_./((a_+b_.*x_^n_)*(c_+d_.*x_^n_)),x_Symbol] :=
  b/(b*c-a*d)*Int[(e*x)^m/(a+b*x^n),x] - d/(b*c-a*d)*Int[(e*x)^m/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,e,n,m},x] && NeQ[b*c-a*d,0]
```

12: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0 \wedge p+2\in\mathbb{Z}^+ \wedge q+2\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.4.12: If $b\,c-a\,d\neq 0 \wedge p+2\in\mathbb{Z}^+ \wedge q+2\in\mathbb{Z}^+$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx \to \int \text{ExpandIntegrand}\left[(e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q,\,x\right] dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  Int[ExpandIntegrand[(e*x)^m*(a+b*x^n)^p*(c+d*x^n)^q,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[b*c-a*d,0] && IGtQ[p,-2] && (IGtQ[q,-2] || EqQ[q,-3] && IntegerQ[(m-1)/2])
```

A. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0 \wedge m\neq -1 \wedge m\neq n-1$

1: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx$ when $b\,c-a\,d\neq 0 \wedge m\neq -1 \wedge m\neq n-1 \wedge (p\in\mathbb{Z} \vee a>0) \wedge (q\in\mathbb{Z} \vee c>0)$

Rule 1.1.3.4.A.1: If $b\,c-a\,d\neq 0 \wedge m\neq -1 \wedge m\neq n-1 \wedge (p\in\mathbb{Z} \vee a>0) \wedge (q\in\mathbb{Z} \vee c>0)$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \,dx \to \frac{a^p\,c^q\,(e\,x)^{m+1}}{e\,(m+1)}\,\text{AppellF1}\left[\frac{m+1}{n},\,-p,\,-q,\,1+\frac{m+1}{n},\,-\frac{b\,x^n}{a},\,-\frac{d\,x^n}{c}\right]$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  a^p*c^q*(e*x)^(m+1)/(e*(m+1))*AppellF1[(m+1)/n,-p,-q,1+(m+1)/n,-b*x^n/a,-d*x^n/c] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && NeQ[b*c-a*d,0] && NeQ[m,-1] && NeQ[m,n-1] &&
  (IntegerQ[p] || GtQ[a,0]) && (IntegerQ[q] || GtQ[c,0])
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \, dx$ when $b\,c-a\,d\neq 0\ \land\ m\neq -1\ \land\ m\neq n-1\ \land\ \neg\,(p\in\mathbb{Z}\ \lor\ a>0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+b\,x^n)^p}{\left(1+\frac{b\,x^n}{a}\right)^p} == 0$

Rule 1.1.3.4.A.2: If $b\,c-a\,d\neq 0\ \land\ m\neq -1\ \land\ m\neq n-1\ \land\ \neg\,(p\in\mathbb{Z}\ \lor\ a>0)$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \, dx \longrightarrow \frac{a^{\text{IntPart}[p]} \left(a+b\,x^n\right)^{\text{FracPart}[p]}}{\left(1+\frac{b\,x^n}{a}\right)^{\text{FracPart}[p]}} \int (e\,x)^m \left(1+\frac{b\,x^n}{a}\right)^p \left(c+d\,x^n\right)^q \, dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^n)^FracPart[p]/(1+b*x^n/a)^FracPart[p]*Int[(e*x)^m*(1+b*x^n/a)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && NeQ[b*c-a*d,0] && NeQ[m,-1] && NeQ[m,n-1] && Not[IntegerQ[p] || GtQ[a,0]]
```

S. $\int u^m (a+b v^n)^p (c+d v^n)^q \, dx$ when $v == e+f x \land u == g v$

1: $\int x^m (a+b v^n)^p (c+d v^n)^q \, dx$ when $v == e+f x \land m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}$, then $x^m F[e+f x] == \frac{1}{f^{m+1}} \text{Subst}[(x-e)^m F[x], x, e+f x] \, \partial_x (e+f x)$

Rule 1.1.3.4.S.1: If $v == e+f x \land m \in \mathbb{Z}$, then

$$\int x^m (a+b v^n)^p (c+d v^n)^q \, dx \to \frac{1}{f^{m+1}} \text{Subst}\left[\int (x-e)^m (a+b x^n)^p (c+d x^n)^q \, dx, x, v\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*v_^n_)^p_.*(c_.+d_.*v_^n_)^q_.,x_Symbol] :=
  1/Coefficient[v,x,1]^(m+1)*Subst[Int[SimplifyIntegrand[(x-Coefficient[v,x,0])^m*(a+b*x^n)^p*(c+d*x^n)^q,x],x],x,v] /;
FreeQ[{a,b,c,d,n,p,q},x] && LinearQ[v,x] && IntegerQ[m] && NeQ[v,x]
```

2: $\int u^m (a + b v^n)^p (c + d v^n)^q \, dx$ when $v == e + f x \land u == g v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u == g v$, then $\partial_x \frac{u^m}{v^m} == 0$

Rule 1.1.3.4.S.2: If $v == e + f x \land u == g v$, then

$$\int u^m (a + b v^n)^p (c + d v^n)^q \, dx \longrightarrow \frac{u^m}{f v^m} \text{Subst}\left[\int x^m (a + b x^n)^p (c + d x^n)^q \, dx, x, v\right]$$

Program code:

```
Int[u_^m_.*(a_.+b_.*v_^n_)^p_.*(c_.+d_.*v_^n_)^q_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q,x],x,v] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && LinearPairQ[u,v,x]
```

N. $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \,dx$

1. $\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \,dx$

1: $\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \,dx$ when $q \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $q \in \mathbb{Z}$, then $\left(c+d\,x^{-n}\right)^q == x^{-n\,q} \left(d+c\,x^n\right)^q$

Rule 1.1.3.4.N.1.1: If $q \in \mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \,dx \longrightarrow \int x^{m-n\,q} \left(a+b\,x^n\right)^p \left(d+c\,x^n\right)^q \,dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.,x_Symbol] :=
  Int[x^(m-n*q)*(a+b*x^n)^p*(d+c*x^n)^q,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[mn,-n] && IntegerQ[q] && (PosQ[n] || Not[IntegerQ[p]])
```

2: $\int x^m \left(a+b x^n\right)^p \left(c+d x^{-n}\right)^q \, dx$ when $q \notin \mathbb{Z} \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n q} \left(c+d x^{-n}\right)^q}{\left(d+c x^n\right)^q} == 0$

Basis: $\frac{x^{n q} \left(c+d x^{-n}\right)^q}{\left(d+c x^n\right)^q} == \frac{x^{n \, \text{FracPart}[q]} \left(c+d x^{-n}\right)^{\text{FracPart}[q]}}{\left(d+c x^n\right)^{\text{FracPart}[q]}}$

Rule 1.1.3.4.N.1.2: If $q \notin \mathbb{Z} \land p \notin \mathbb{Z}$, then

$$\int x^m \left(a+b x^n\right)^p \left(c+d x^{-n}\right)^q \, dx \longrightarrow \frac{x^{n \, \text{FracPart}[q]} \left(c+d x^{-n}\right)^{\text{FracPart}[q]}}{\left(d+c x^n\right)^{\text{FracPart}[q]}} \int x^{m-n q} \left(a+b x^n\right)^p \left(d+c x^n\right)^q \, dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_,x_Symbol] :=
  x^(n*FracPart[q])*(c+d*x^(-n))^FracPart[q]/(d+c*x^n)^FracPart[q]*Int[x^(m-n*q)*(a+b*x^n)^p*(d+c*x^n)^q,x] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && EqQ[mn,-n] && Not[IntegerQ[q]] && Not[IntegerQ[p]]
```

2: $\int (e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^{-n}\right)^q\,dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Basis: $\frac{(e\,x)^m}{x^m} == \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.4.N.2:

$$\int (e\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^{-n}\right)^q\,dx \longrightarrow \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^{-n}\right)^q\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(a_.+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^(-n))^q,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && EqQ[mn,-n]
```

```
(* IntBinomialQ[a,b,c,d,e,m,n,p,q,x] returns True iff (e*x)^m*(a+b*x^n)^p*(c+d*x^n)^q is integrable wrt x in terms of non-Appell funct
IntBinomialQ[a_,b_,c_,d_,e_,m_,n_,p_,q_,x_Symbol] :=
  IntegersQ[p,q] || IGtQ[p,0] || IGtQ[q,0] ||
  EqQ[n,2] && (IntegersQ[m,2*p,2*q] || IntegersQ[2*m,p,2*q] || IntegersQ[2*m,2*p,q]) ||
  EqQ[n,4] && (IntegersQ[m,p,2*q] || IntegersQ[m,2*p,q]) ||
  EqQ[n,2] && IntegersQ[m/2,p+1/3,q] && (EqQ[b*c+3*a*d,0] || EqQ[b*c-9*a*d,0]) ||
  EqQ[n,2] && IntegersQ[m/2,q+1/3,p] && (EqQ[a*d+3*b*c,0] || EqQ[a*d-9*b*c,0]) ||
  EqQ[n,3] && IntegersQ[(m-1)/3,q,p-1/2] && (EqQ[b*c-4*a*d,0] || EqQ[b*c+8*a*d,0] || EqQ[b^2*c^2-20*a*b*c*d-8*a^2*d^2,0]) ||
  EqQ[n,3] && IntegersQ[(m-1)/3,p,q-1/2] && (EqQ[4*b*c-a*d,0] || EqQ[8*b*c+a*d,0] || EqQ[8*b^2*c^2+20*a*b*c*d-a^2*d^2,0])
```

Rules for integrands of the form $u\left(a_1+b_1 x^{n/2}\right)^p\left(a_2+b_2 x^{n/2}\right)^p F[x^n]$

1: $\int u\left(a_1+b_1 x^{n/2}\right)^p\left(a_2+b_2 x^{n/2}\right)^p F[x^n]\,dx$ when $a_2 b_1+a_1 b_2==0\land\left(p\in\mathbb{Z}\lor\left(a_1>0\land a_2>0\right)\right)$

Derivation: Algebraic simplification

Basis: If $a_2 b_1+a_1 b_2==0\land\left(p\in\mathbb{Z}\lor\left(a_1>0\land a_2>0\right)\right)$, then $\left(a_1+b_1 x^{n/2}\right)^p\left(a_2+b_2 x^{n/2}\right)^p==\left(a_1 a_2+b_1 b_2 x^n\right)^p$

Rule: If $a_2 b_1+a_1 b_2==0\land\left(p\in\mathbb{Z}\lor\left(a_1>0\land a_2>0\right)\right)$, then

$$\int u\left(a_1+b_1 x^{n/2}\right)^p\left(a_2+b_2 x^{n/2}\right)^p F[x^n]\,dx\longrightarrow\int u\left(a_1 a_2+b_1 b_2 x^n\right)^p F[x^n]\,dx$$

Program code:

```
Int[u_.*(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_.)^q_.,x_Symbol] :=
  Int[u*(a1*a2+b1*b2*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a1,b1,a2,b2,c,d,n,p,q},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && (IntegerQ[p] || GtQ[a1,0] && GtQ[a2,0])
```

```
Int[u_.*(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_.+e_.*x_^n2_.)^q_.,x_Symbol] :=
  Int[u*(a1*a2+b1*b2*x^n)^p*(c+d*x^n+e*x^(2*n))^q,x] /;
FreeQ[{a1,b1,a2,b2,c,d,e,n,p,q},x] && EqQ[non2,n/2] && EqQ[n2,2*n] && EqQ[a2*b1+a1*b2,0] && (IntegerQ[p] || GtQ[a1,0] && GtQ[a2,0])
```

2: $\int u\left(a_1+b_1 x^{n/2}\right)^p\left(a_2+b_2 x^{n/2}\right)^p F\left[x^n\right] dx$ when $a_2 b_1+a_1 b_2==0 \land \neg\left(p\in\mathbb{Z} \lor \left(a_1>0\land a_2>0\right)\right)$

Derivation: Piecewise constant extraction

Basis: If $a_2 b_1+a_1 b_2==0$, then $\partial_x \frac{\left(a_1+b_1 x^{n/2}\right)^p\left(a_2+b_2 x^{n/2}\right)^p}{\left(a_1 a_2+b_1 b_2 x^n\right)^p}==0$

Rule: If $a_2 b_1+a_1 b_2==0$, then

$$\int u\left(a_1+b_1 x^{n/2}\right)^p\left(a_2+b_2 x^{n/2}\right)^p F\left[x^n\right] dx \to \frac{\left(a_1+b_1 x^{n/2}\right)^{\text{FracPart}[p]}\left(a_2+b_2 x^{n/2}\right)^{\text{FracPart}[p]}}{\left(a_1 a_2+b_1 b_2 x^n\right)^{\text{FracPart}[p]}}\int u\left(a_1 a_2+b_1 b_2 x^n\right)^p F\left[x^n\right] dx$$

Program code:

```
Int[u_.*(a1_+b1_.*x_^non2_.)^p_*(a2_+b2_.*x_^non2_.)^p_*(c_+d_.*x_^n_.)^q_.,x_Symbol] :=
  (a1+b1*x^(n/2))^FracPart[p]*(a2+b2*x^(n/2))^FracPart[p]/(a1*a2+b1*b2*x^n)^FracPart[p]*
    Int[u*(a1*a2+b1*b2*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a1,b1,a2,b2,c,d,n,p,q},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0]
```

```
Int[u_.*(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_.+e_.*x_^n2_.)^q_.,x_Symbol] :=
  (a1+b1*x^(n/2))^FracPart[p]*(a2+b2*x^(n/2))^FracPart[p]/(a1*a2+b1*b2*x^n)^FracPart[p]*
    Int[u*(a1*a2+b1*b2*x^n)^p*(c+d*x^n+e*x^(2*n))^q,x] /;
FreeQ[{a1,b1,a2,b2,c,d,e,n,p,q},x] && EqQ[non2,n/2] && EqQ[n2,2*n] && EqQ[a2*b1+a1*b2,0]
```