Rules for integrands of the form $\text{Trig}[c + d\,x]^m\,(a\,\text{Cos}[c + d\,x] + b\,\text{Sin}[c + d\,x])^n$

1. $\int (a\,\text{Cos}[c + d\,x] + b\,\text{Sin}[c + d\,x])^n\,dx$

   1: $\int (a\,\text{Cos}[c + d\,x] + b\,\text{Sin}[c + d\,x])^n\,dx$ when $a^2 + b^2 == 0$

Reference: Integration by substitution

Basis: If $a^2 + b^2 == 0$, then

$(a\,\text{Cos}[c + d\,x] + b\,\text{Sin}[c + d\,x])^n == \frac{a\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-1}}{b\,d}\,\partial_x\,(a\,\text{Cos}[c + d\,x] + b\,\text{Sin}[c + d\,x])$

Rule: If $a^2 + b^2 == 0$, then

$$\int (a\,\text{Cos}[c + d\,x] + b\,\text{Sin}[c + d\,x])^n\,dx \longrightarrow \frac{a\,(a\,\text{Cos}[c + d\,x] + b\,\text{Sin}[c + d\,x])^n}{b\,d\,n}$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(b*d*n) /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2+b^2,0]
```

2. $\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0$

1. $\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\,\land\,n>1$

1: $\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\,\land\,\frac{n-1}{2}\in\mathbb{Z}^+$

Reference: G&R 2.557'

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then

$$(a\,\text{Cos}[z]+b\,\text{Sin}[z])^n == -\left(a^2+b^2-(b\,\text{Cos}[z]-a\,\text{Sin}[z])^2\right)^{\frac{n-1}{2}}\partial_z\,(b\,\text{Cos}[z]-a\,\text{Sin}[z])$$

Rule: If $a^2+b^2\neq 0\,\land\,\frac{n-1}{2}\in\mathbb{Z}^+$, then

$$\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \to -\frac{1}{d}\,\text{Subst}\left[\int \left(a^2+b^2-x^2\right)^{\frac{n-1}{2}}dx,\,x,\,b\,\text{Cos}[c+d\,x]-a\,\text{Sin}[c+d\,x]\right]$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -1/d*Subst[Int[(a^2+b^2-x^2)^((n-1)/2),x],x,b*Cos[c+d*x]-a*Sin[c+d*x]] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && IGtQ[(n-1)/2,0]
```

2: $\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\,\land\,\frac{n-1}{2}\notin\mathbb{Z}\,\land\,n>1$

Derivation: Integration by parts with a double-back flip

Rule: If $a^2+b^2\neq 0\,\land\,\frac{n-1}{2}\notin\mathbb{Z}\,\land\,n>1$, then

$$\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \to$$

$$-\frac{(b\,\text{Cos}[c+d\,x]-a\,\text{Sin}[c+d\,x])\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-1}}{d\,n}+\frac{(n-1)\,(a^2+b^2)}{n}\int(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-2}\,dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*Cos[c+d*x]-a*Sin[c+d*x])*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/(d*n) +
  (n-1)*(a^2+b^2)/n*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && Not[IntegerQ[(n-1)/2]] && GtQ[n,1]
```

2. $\int(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\land n\leq -1$

1: $\int\frac{1}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0$

Reference: G&R 2.557'

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then

$(a\,\text{Cos}[z]+b\,\text{Sin}[z])^n == -\left(a^2+b^2-(b\,\text{Cos}[z]-a\,\text{Sin}[z])^2\right)^{\frac{n-1}{2}}\,\partial_z\,(b\,\text{Cos}[z]-a\,\text{Sin}[z])$

Rule: If $a^2+b^2\neq 0$, then

$$\int\frac{1}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx\to-\frac{1}{d}\,\text{Subst}\left[\int\frac{1}{a^2+b^2-x^2}\,dx,\,x,\,b\,\text{Cos}[c+d\,x]-a\,\text{Sin}[c+d\,x]\right]$$

Program code:

```
Int[1/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -1/d*Subst[Int[1/(a^2+b^2-x^2),x],x,b*Cos[c+d*x]-a*Sin[c+d*x]] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

2: $\int \frac{1}{(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^2}\,dx$ when $a^2+b^2\neq 0$

Reference: G&R 2.557.5b'

Rule: If $a^2+b^2\neq 0$, then

$$\int \frac{1}{(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^2}\,dx \rightarrow \frac{\mathrm{Sin}[c+d\,x]}{a\,d\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])}$$

Program code:

```
Int[1/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^2,x_Symbol] :=
  Sin[c+d*x]/(a*d*(a*Cos[c+d*x]+b*Sin[c+d*x])) /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

3: $\int (a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\wedge n<-1\wedge n\neq -2$

Derivation: Integration by parts with a double-back flip

Rule: If $a^2+b^2\neq 0\wedge n<-1\wedge n\neq -2$, then

$$\int (a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx \rightarrow$$
$$\frac{(b\,\mathrm{Cos}[c+d\,x]-a\,\mathrm{Sin}[c+d\,x])\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^{n+1}}{d\,(n+1)\,(a^2+b^2)}+\frac{n+2}{(n+1)\,(a^2+b^2)}\int (a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^{n+2}\,dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*Cos[c+d*x]-a*Sin[c+d*x])*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1)/(d*(n+1)*(a^2+b^2)) +
  (n+2)/((n+1)*(a^2+b^2))*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1] && NeQ[n,-2]
```

3. $\int (a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n \, dx$ when $a^2+b^2 \neq 0 \wedge \neg (n \geq 1 \vee n \leq -1)$

1: $\int (a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n \, dx$ when $\neg (n \geq 1 \vee n \leq -1) \wedge a^2+b^2>0$

Derivation: Algebraic simplification

Basis: If $a^2+b^2 \neq 0$, then $a \operatorname{Cos}[z]+b \operatorname{Sin}[z] = \sqrt{a^2+b^2} \operatorname{Cos}[z-\operatorname{ArcTan}[a, b]]$

Rule: If $\neg (n \geq 1 \vee n \leq -1) \wedge a^2+b^2>0$, then

$$\int (a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n \, dx \to \left(a^2+b^2\right)^{n/2} \int (\operatorname{Cos}[c+d x-\operatorname{ArcTan}[a, b]])^n \, dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (a^2+b^2)^(n/2)*Int[(Cos[c+d*x-ArcTan[a,b]])^n,x] /;
FreeQ[{a,b,c,d,n},x] && Not[GeQ[n,1] || LeQ[n,-1]] && GtQ[a^2+b^2,0]
```

2: $\int (a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n \, dx$ when $\neg (n \geq 1 \vee n \leq -1) \wedge \neg \left(a^2+b^2 \geq 0\right)$

Derivation: Piecewise constant extraction and algebraic simplification

Basis: $\partial_x \frac{(a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n}{\left(\frac{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}{\sqrt{a^2+b^2}}\right)^n} = 0$

Basis: If $a^2+b^2 \neq 0$, then $\frac{a \operatorname{Cos}[z]+b \operatorname{Sin}[z]}{\sqrt{a^2+b^2}} = \operatorname{Cos}[z-\operatorname{ArcTan}[a, b]]$

Rule: If $\neg (n \geq 1 \vee n \leq -1) \wedge \neg \left(a^2+b^2 \geq 0\right)$, then

$$\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \longrightarrow \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\left(\frac{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}{\sqrt{a^2+b^2}}\right)^n}\int \left(\frac{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}{\sqrt{a^2+b^2}}\right)^n dx$$

$$\longrightarrow \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\left(\frac{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}{\sqrt{a^2+b^2}}\right)^n}\int (\text{Cos}[c+d\,x-\text{ArcTan}[a,b]])^n\,dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (a*Cos[c+d*x]+b*Sin[c+d*x])^n/((a*Cos[c+d*x]+b*Sin[c+d*x])/Sqrt[a^2+b^2])^n*Int[Cos[c+d*x-ArcTan[a,b]]^n,x] /;
FreeQ[{a,b,c,d,n},x] && Not[GeQ[n,1] || LeQ[n,-1]] && Not[GtQ[a^2+b^2,0] || EqQ[a^2+b^2,0]]
```

2. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$

   1. $\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]^n}\,dx$ when $n\in\mathbb{Z}$

      1. $\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]^n}\,dx$ when $n\in\mathbb{Z}\land a^2+b^2==0$

         1: $\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]^n}\,dx$ when $a^2+b^2==0\land n>1$

Note: Compare this with the rule for integrands of the form $(a+b\,\text{Cot}[c+d\,x])^n$ when $a^2+b^2==0\land n>1$.

Rule: If $a^2+b^2==0\land n>1$, then

$$\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]^n}\,dx \longrightarrow -\frac{a\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-1}}{d\,(n-1)\,\text{Sin}[c+d\,x]^{n-1}}+2\,b\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-1}}{\text{Sin}[c+d\,x]^{n-1}}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -a*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/(d*(n-1)*Sin[c+d*x]^(n-1)) +
  2*b*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/Sin[c+d*x]^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && GtQ[n,1]
```

```
Int[cos[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  b*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/(d*(n-1)*Cos[c+d*x]^(n-1)) +
  2*a*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/Cos[c+d*x]^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && GtQ[n,1]
```

2: $\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]^n}\,dx$ when $a^2+b^2==0 \land n<0$

Note: Compare this with the rule for integrands of the form $(a+b\,\text{Cot}[c+d\,x])^n$ when $a^2+b^2==0 \land n<0$.

Rule: If $a^2+b^2==0 \land n<0$, then

$$\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]^n}\,dx \longrightarrow \frac{a\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{2\,b\,d\,n\,\text{Sin}[c+d\,x]^n}+\frac{1}{2\,b}\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n+1}}{\text{Sin}[c+d\,x]^{n+1}}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(2*b*d*n*Sin[c+d*x]^n) +
  1/(2*b)*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1)/Sin[c+d*x]^(n+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && LtQ[n,0]
```

```
Int[cos[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(2*a*d*n*Cos[c+d*x]^n) +
  1/(2*a)*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1)/Cos[c+d*x]^(n+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && LtQ[n,0]
```

3: $\int \frac{(a \cos[c+d x]+b \sin[c+d x])^n}{\sin[c+d x]^n} \, dx$ when $a^2+b^2==0 \land n \notin \mathbb{Z}$

Rule: If $a^2+b^2==0 \land n \notin \mathbb{Z}$, then

$$\int \frac{(a \cos[c+d x]+b \sin[c+d x])^n}{\sin[c+d x]^n} \, dx \longrightarrow \frac{a\,(a \cos[c+d x]+b \sin[c+d x])^n}{2\,b\,d\,n \sin[c+d x]^n} \text{Hypergeometric2F1}\left[1,\,n,\,n+1,\,\frac{b+a \cot[c+d x]}{2\,b}\right]$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(2*b*d*n*Sin[c+d*x]^n)*Hypergeometric2F1[1,n,n+1,(b+a*Cot[c+d*x])/(2*b)] /;
FreeQ[{a,b,c,d,n},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && Not[IntegerQ[n]]
```

```
Int[cos[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(2*a*d*n*Cos[c+d*x]^n)*Hypergeometric2F1[1,n,n+1,(a+b*Tan[c+d*x])/(2*a)] /;
FreeQ[{a,b,c,d,n},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && Not[IntegerQ[n]]
```

2: $\int \frac{(a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n}{\operatorname{Sin}[c+d x]^n} \, dx$ when $n \in \mathbb{Z} \wedge a^2+b^2 \neq 0$

Derivation: Algebraic simplification

Basis: $\frac{a \operatorname{Cos}[z]+b \operatorname{Sin}[z]}{\operatorname{Sin}[z]} == b+a \operatorname{Cot}[z]$

Rule: If $n \in \mathbb{Z} \wedge a^2+b^2 \neq 0$, then

$$\int \frac{(a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n}{\operatorname{Sin}[c+d x]^n} \, dx \longrightarrow \int (b+a \operatorname{Cot}[c+d x])^n \, dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  Int[(b+a*Cot[c+d*x])^n,x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && IntegerQ[n] && NeQ[a^2+b^2,0]
```

```
Int[cos[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  Int[(a+b*Tan[c+d*x])^n,x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && IntegerQ[n] && NeQ[a^2+b^2,0]
```

2: $\int \operatorname{Sin}[c+d x]^m (a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n \, dx$ when $n \in \mathbb{Z} \wedge \frac{m+n}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $\operatorname{Sin}[c+d x]^m (a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n == \operatorname{Sin}[c+d x]^{m+n} \frac{(a+b \operatorname{Tan}[c+d x])^n}{\operatorname{Tan}[c+d x]^n}$

Basis: If $\frac{m+n}{2} \in \mathbb{Z}$, then $\operatorname{Sin}[c+d x]^{m+n} \frac{(a+b \operatorname{Tan}[c+d x])^n}{\operatorname{Tan}[c+d x]^n} == \frac{1}{d} \frac{\operatorname{Tan}[c+d x]^m (a+b \operatorname{Tan}[c+d x])^n}{\left(1+\operatorname{Tan}[c+d x]^2\right)^{\frac{m+n+2}{2}}} \partial_x \operatorname{Tan}[c+d x]$

Rule: If $n \in \mathbb{Z} \wedge \frac{m+n}{2} \in \mathbb{Z}$, then

$$\int \mathrm{Sin}[c+d\,x]^m\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx \to \frac{1}{d}\,\mathrm{Subst}\Big[\int \frac{x^m\,(a+b\,x)^n}{(1+x^2)^{\frac{m+n+2}{2}}}\,dx,\,x,\,\mathrm{Tan}[c+d\,x]\Big]$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  1/d*Subst[Int[x^m*(a+b*x)^n/(1+x^2)^((m+n+2)/2),x],x,Tan[c+d*x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[n] && IntegerQ[(m+n)/2] && NeQ[n,-1] && Not[GtQ[n,0] && GtQ[m,1]]
```

```
Int[cos[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -1/d*Subst[Int[x^m*(b+a*x)^n/(1+x^2)^((m+n+2)/2),x],x,Cot[c+d*x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[n] && IntegerQ[(m+n)/2] && NeQ[n,-1] && Not[GtQ[n,0] && GtQ[m,1]]
```

3: $\int \mathrm{Sin}[c+d\,x]^m\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx$ when $m \in \mathbb{Z}\ \land\ n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}\ \land\ n \in \mathbb{Z}^+$, then

$$\int \mathrm{Sin}[c+d\,x]^m\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx \to \int \mathrm{ExpandTrig}\big[\mathrm{Sin}[c+d\,x]^m\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n,\,x\big]\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  Int[ExpandTrig[sin[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && IGtQ[n,0]
```

```
Int[cos[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  Int[ExpandTrig[cos[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && IGtQ[n,0]
```

4: $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2==0\,\land\,n\in\mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $a^2+b^2==0$, then $a\,\text{Cos}[z]+b\,\text{Sin}[z]==a\,b\,(b\,\text{Cos}[z]+a\,\text{Sin}[z])^{-1}$

Rule: If $a^2+b^2==0\,\land\,n\in\mathbb{Z}^-$, then

$$\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \longrightarrow a^n\,b^n\int \text{Sin}[c+d\,x]^m\,(b\,\text{Cos}[c+d\,x]+a\,\text{Sin}[c+d\,x])^{-n}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a^n*b^n*Int[Sin[c+d*x]^m*(b*Cos[c+d*x]+a*Sin[c+d*x])^(-n),x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[a^2+b^2,0] && ILtQ[n,0]
```

```
Int[cos[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a^n*b^n*Int[Cos[c+d*x]^m*(b*Cos[c+d*x]+a*Sin[c+d*x])^(-n),x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[a^2+b^2,0] && ILtQ[n,0]
```

5. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0$

 1. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\land n>0$

  2. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\land n>1$

   1. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\land n>1\land m>0$

   2. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\land n>1\land m<0$

    1: $\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0\land n>1$

Derivation: Algebraic expansion and power rule for integration

Basis: $\frac{(a\,\text{Cos}[z]+b\,\text{Sin}[z])^2}{\text{Sin}[z]} == a\,(b\,\text{Cos}[z]-a\,\text{Sin}[z])+b\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])+\frac{a^2}{\text{Sin}[z]}$

Rule: If $a^2+b^2\neq 0\land n<-1$, then

$$\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]}\,dx\,\rightarrow$$

$$\frac{a\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-1}}{d\,(n-1)}+b\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-1}\,dx+a^2\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-2}}{\text{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/sin[c_.+d_.*x_],x_Symbol] :=
  a*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/(d*(n-1)) +
  b*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1),x] +
  a^2*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2)/Sin[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/cos[c_.+d_.*x_],x_Symbol] :=
  -b*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/(d*(n-1)) +
  a*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1),x] +
  b^2*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2)/Cos[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

2: $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\,\wedge\,n>1\,\wedge\,m<-1$

Derivation: Algebraic expansion

Basis: $(a\,\text{Cos}[z]+b\,\text{Sin}[z])^2 == -(a^2+b^2)\,\text{Sin}[z]^2+2\,b\,\text{Sin}[z]\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])+a^2$

Rule: If $a^2+b^2\neq 0\,\wedge\,n>1\,\wedge\,m<-1$, then

$$\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \longrightarrow$$
$$-(a^2+b^2)\int \text{Sin}[c+d\,x]^{m+2}\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-2}\,dx+$$
$$2\,b\int \text{Sin}[c+d\,x]^{m+1}\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-1}\,dx+a^2\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n-2}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(a^2+b^2)*Int[Sin[c+d*x]^(m+2)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2),x] +
  2*b*Int[Sin[c+d*x]^(m+1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1),x] +
  a^2*Int[Sin[c+d*x]^m*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && GtQ[n,1] && LtQ[m,-1]
```

```
Int[cos[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(a^2+b^2)*Int[Cos[c+d*x]^(m+2)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2),x] +
  2*a*Int[Cos[c+d*x]^(m+1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1),x] +
  b^2*Int[Cos[c+d*x]^m*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && GtQ[n,1] && LtQ[m,-1]
```

2. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\wedge n<0$

   1. $\int \frac{\text{Sin}[c+d\,x]^m}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0$

      1. $\int \frac{\text{Sin}[c+d\,x]^m}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0\wedge m>0$

         1: $\int \frac{\text{Sin}[c+d\,x]}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\text{Sin}[z]}{a\,\text{Cos}[z]+b\,\text{Sin}[z]}==\frac{b}{a^2+b^2}-\frac{a\,(b\,\text{Cos}[z]-a\,\text{Sin}[z])}{(a^2+b^2)\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$

Rule: If $a^2+b^2\neq 0$, then

$$\int \frac{\text{Sin}[c+d\,x]}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx\;\to\;\frac{b\,x}{a^2+b^2}-\frac{a}{a^2+b^2}\int \frac{b\,\text{Cos}[c+d\,x]-a\,\text{Sin}[c+d\,x]}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  b*x/(a^2+b^2) -
  a/(a^2+b^2)*Int[(b*Cos[c+d*x]-a*Sin[c+d*x])/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

```
Int[cos[c_.+d_.*x_]/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  a*x/(a^2+b^2) +
  b/(a^2+b^2)*Int[(b*Cos[c+d*x]-a*Sin[c+d*x])/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

2: $\int \frac{\text{Sin}[c+d\,x]^m}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0\land m>1$

Derivation: Algebraic expansion and power rule for integration

Basis: $\frac{\text{Sin}[z]^2}{a\,\text{Cos}[z]+b\,\text{Sin}[z]} == -\frac{a\,\text{Cos}[z]}{a^2+b^2}+\frac{b\,\text{Sin}[z]}{a^2+b^2}+\frac{a^2}{(a^2+b^2)\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$

Rule: If $a^2+b^2\neq 0\land m>1$, then

$$\int \frac{\text{Sin}[c+d\,x]^m}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \to -\frac{a\,\text{Sin}[c+d\,x]^{m-1}}{d\,(a^2+b^2)\,(m-1)}+\frac{b}{a^2+b^2}\int \text{Sin}[c+d\,x]^{m-1}\,dx+\frac{a^2}{a^2+b^2}\int \frac{\text{Sin}[c+d\,x]^{m-2}}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -a*Sin[c+d*x]^(m-1)/(d*(a^2+b^2)*(m-1)) +
  b/(a^2+b^2)*Int[Sin[c+d*x]^(m-1),x] +
  a^2/(a^2+b^2)*Int[Sin[c+d*x]^(m-2)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && GtQ[m,1]
```

```
Int[cos[c_.+d_.*x_]^m_/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  b*Cos[c+d*x]^(m-1)/(d*(a^2+b^2)*(m-1)) +
  a/(a^2+b^2)*Int[Cos[c+d*x]^(m-1),x] +
  b^2/(a^2+b^2)*Int[Cos[c+d*x]^(m-2)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && GtQ[m,1]
```

2. $\int \frac{\text{Sin}[c+d\,x]^m}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0\land m<0$

1: $\int \frac{1}{\text{Sin}[c+d\,x]\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])}\,dx$ when $a^2+b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\text{Sin}[z]\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}==\frac{\text{Cot}[z]}{a}-\frac{b\,\text{Cos}[z]-a\,\text{Sin}[z]}{a\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$

Rule: If $a^2+b^2\neq 0$, then

$$\int \frac{1}{\text{Sin}[c+d\,x]\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])}\,dx\to\frac{1}{a}\int \text{Cot}[c+d\,x]\,dx-\frac{1}{a}\int \frac{b\,\text{Cos}[c+d\,x]-a\,\text{Sin}[c+d\,x]}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[1/(sin[c_.+d_.*x_]*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  1/a*Int[Cot[c+d*x],x] -
  1/a*Int[(b*Cos[c+d*x]-a*Sin[c+d*x])/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

```
Int[1/(cos[c_.+d_.*x_]*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  1/b*Int[Tan[c+d*x],x] +
  1/b*Int[(b*Cos[c+d*x]-a*Sin[c+d*x])/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

2: $\int \frac{\text{Sin}[c+d\,x]^m}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0\land m<-1$

Derivation: Algebraic expansion and power rule for integration

Basis: $\frac{1}{a\,\text{Cos}[z]+b\,\text{Sin}[z]}==\frac{\text{Cos}[z]}{a}-\frac{b\,\text{Sin}[z]}{a^2}+\frac{(a^2+b^2)\,\text{Sin}[z]^2}{a^2\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$

Rule: If $a^2+b^2\neq 0 \land m<-1$, then

$$\int \frac{\text{Sin}[c+d\,x]^m}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \to \frac{\text{Sin}[c+d\,x]^{m+1}}{a\,d\,(m+1)}-\frac{b}{a^2}\int \text{Sin}[c+d\,x]^{m+1}\,dx+\frac{a^2+b^2}{a^2}\int \frac{\text{Sin}[c+d\,x]^{m+2}}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  Sin[c+d*x]^(m+1)/(a*d*(m+1)) -
  b/a^2*Int[Sin[c+d*x]^(m+1),x] +
  (a^2+b^2)/a^2*Int[Sin[c+d*x]^(m+2)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[m,-1]
```

```
Int[cos[c_.+d_.*x_]^m_/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -Cos[c+d*x]^(m+1)/(b*d*(m+1)) -
  a/b^2*Int[Cos[c+d*x]^(m+1),x] +
  (a^2+b^2)/b^2*Int[Cos[c+d*x]^(m+2)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[m,-1]
```

2. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0 \land n<-1$

   1. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0 \land n<-1 \land m>0$

   2. $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0 \land n<-1 \land m<0$

      1: $\int \frac{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0 \land n<-1$

Derivation: Algebraic expansion and power rule for integration

Basis: $\frac{1}{\text{Sin}[z]} = -\frac{(b\,\text{Cos}[z]-a\,\text{Sin}[z])}{a}-\frac{b\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}{a^2}+\frac{(a\,\text{Cos}[z]+b\,\text{Sin}[z])^2}{a^2\,\text{Sin}[z]}$

Rule: If $a^2+b^2\neq 0 \land n<-1$, then

$$\int \frac{(a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^n}{\operatorname{Sin}[c+d x]} \, dx \longrightarrow$$

$$-\frac{(a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^{n+1}}{a d (n+1)}-\frac{b}{a^2} \int (a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^{n+1} \, dx+\frac{1}{a^2} \int \frac{(a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x])^{n+2}}{\operatorname{Sin}[c+d x]} \, dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/sin[c_.+d_.*x_],x_Symbol] :=
  -(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1)/(a*d*(n+1)) -
  b/a^2*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1),x] +
  1/a^2*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+2)/Sin[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/cos[c_.+d_.*x_],x_Symbol] :=
  (a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1)/(b*d*(n+1)) -
  a/b^2*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1),x] +
  1/b^2*Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+2)/Cos[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

2: $\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2\neq 0\,\land\,n<-1\,\land\,m<-1$

Derivation: Algebraic expansion

Basis: $1 == \frac{(a^2+b^2)\,\text{Sin}[z]^2}{a^2} - \frac{2\,b\,\text{Sin}[z]\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}{a^2} + \frac{(a\,\text{Cos}[z]+b\,\text{Sin}[z])^2}{a^2}$

Rule: If $a^2+b^2\neq 0\,\land\,n<-1\,\land\,m<-1$, then

$$\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \longrightarrow$$
$$\frac{a^2+b^2}{a^2}\int \text{Sin}[c+d\,x]^{m+2}\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx -$$
$$\frac{2\,b}{a^2}\int \text{Sin}[c+d\,x]^{m+1}\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n+1}\,dx + \frac{1}{a^2}\int \text{Sin}[c+d\,x]^m\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{n+2}\,dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (a^2+b^2)/a^2*Int[Sin[c+d*x]^(m+2)*(a*Cos[c+d*x]+b*Sin[c+d*x])^n,x] -
  2*b/a^2*Int[Sin[c+d*x]^(m+1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1),x] +
  1/a^2*Int[Sin[c+d*x]^m*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1] && LtQ[m,-1]
```

```
Int[cos[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (a^2+b^2)/b^2*Int[Cos[c+d*x]^(m+2)*(a*Cos[c+d*x]+b*Sin[c+d*x])^n,x] -
  2*a/b^2*Int[Cos[c+d*x]^(m+1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1),x] +
  1/b^2*Int[Cos[c+d*x]^m*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1] && LtQ[m,-1]
```

3. $\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx$

1. $\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx$ when $p>0$

1: $\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, then

$$\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx\,\longrightarrow$$
$$\int \text{ExpandTrig}\left[\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p,\,x\right]dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^p_.,x_Symbol] :=
  Int[ExpandTrig[cos[c+d*x]^m*sin[c+d*x]^n*(a*cos[c+d*x]+b*sin[c+d*x])^p,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p,0]
```

2. $\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx$ when $p<0$

1: $\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx$ when $a^2+b^2==0\,\land\,p\in\mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $a^2+b^2==0$, then $a\,\text{Cos}[z]+b\,\text{Sin}[z]==a\,b\,(b\,\text{Cos}[z]+a\,\text{Sin}[z])^{-1}$

Rule: If $a^2+b^2==0\,\land\,p\in\mathbb{Z}^-$, then

$$\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx\,\longrightarrow$$
$$a^p\,b^p\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(b\,\text{Cos}[c+d\,x]+a\,\text{Sin}[c+d\,x])^{-p}\,dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  a^p*b^p*Int[Cos[c+d*x]^m*Sin[c+d*x]^n*(b*Cos[c+d*x]+a*Sin[c+d*x])^(-p),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[a^2+b^2,0] && ILtQ[p,0]
```

2. $\int \frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$

1: $\int \frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $a^2+b^2\neq 0\,\land\,m\in\mathbb{Z}^+\land\,n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\text{Cos}[z]\,\text{Sin}[z]}{a\,\text{Cos}[z]+b\,\text{Sin}[z]}==\frac{b\,\text{Cos}[z]}{a^2+b^2}+\frac{a\,\text{Sin}[z]}{a^2+b^2}-\frac{a\,b}{(a^2+b^2)\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$

Rule: If $a^2+b^2\neq 0\,\land\,m\in\mathbb{Z}^+\land\,n\in\mathbb{Z}^+$, then

$$\int \frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow$$
$$\frac{b}{a^2+b^2}\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^{n-1}\,dx+\frac{a}{a^2+b^2}\int \text{Cos}[c+d\,x]^{m-1}\,\text{Sin}[c+d\,x]^n\,dx-\frac{a\,b}{a^2+b^2}\int \frac{\text{Cos}[c+d\,x]^{m-1}\,\text{Sin}[c+d\,x]^{n-1}}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_./(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  b/(a^2+b^2)*Int[Cos[c+d*x]^m*Sin[c+d*x]^(n-1),x] +
  a/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^n,x] -
  a*b/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^(n-1)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && IGtQ[m,0] && IGtQ[n,0]
```

2: $\int \frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$ when $(m\mid n)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}$, then

$$\int \frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow \int \text{ExpandTrig}\left[\frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]},\,x\right]dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_./(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  Int[ExpandTrig[cos[c+d*x]^m*sin[c+d*x]^n/(a*cos[c+d*x]+b*sin[c+d*x]),x],x] /;
FreeQ[{a,b,c,d,m,n},x] && IntegersQ[m,n]
```

3: $\int \cos[c+d\,x]^m \sin[c+d\,x]^n\,(a\cos[c+d\,x]+b\sin[c+d\,x])^p\,dx$ when $a^2+b^2\neq 0 \land m\in\mathbb{Z}^+ \land m\in\mathbb{Z}^+ \land p\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $\frac{\cos[z]\sin[z]}{a\cos[z]+b\sin[z]} == \frac{b\cos[z]}{a^2+b^2}+\frac{a\sin[z]}{a^2+b^2}-\frac{a\,b}{(a^2+b^2)\,(a\cos[z]+b\sin[z])}$

Rule: If $a^2+b^2\neq 0 \land m\in\mathbb{Z}^+ \land m\in\mathbb{Z}^+ \land p\in\mathbb{Z}^-$, then

$$\int \cos[c+d\,x]^m \sin[c+d\,x]^n\,(a\cos[c+d\,x]+b\sin[c+d\,x])^p\,dx \longrightarrow$$
$$\frac{b}{a^2+b^2}\int \cos[c+d\,x]^m \sin[c+d\,x]^{n-1}\,(a\cos[c+d\,x]+b\sin[c+d\,x])^{p+1}\,dx +$$
$$\frac{a}{a^2+b^2}\int \cos[c+d\,x]^{m-1} \sin[c+d\,x]^{n}\,(a\cos[c+d\,x]+b\sin[c+d\,x])^{p+1}\,dx -$$
$$\frac{a\,b}{a^2+b^2}\int \cos[c+d\,x]^{m-1} \sin[c+d\,x]^{n-1}\,(a\cos[c+d\,x]+b\sin[c+d\,x])^{p}\,dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  b/(a^2+b^2)*Int[Cos[c+d*x]^m*Sin[c+d*x]^(n-1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(p+1),x] +
  a/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^n*(a*Cos[c+d*x]+b*Sin[c+d*x])^(p+1),x] -
  a*b/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^(n-1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && IGtQ[m,0] && IGtQ[n,0] && ILtQ[p,0]
```