Rules for integrands of the form $(d + e x)^m P_q[x] \left(a + b x + c x^2\right)^p$ when $q > 1$

1: $\int (d + e x)^m P_q[x] \left(a + b x + c x^2\right)^p \, dx$ when $\text{PolynomialRemainder}[P_q[x], d + e x, x] = 0$

Derivation: Algebraic simplification

Rule 1.2.1.9.1: If $\text{PolynomialRemainder}[P_q[x], d + e x, x] = 0$, then

$$\int (d + e x)^m P_q[x] \left(a + b x + c x^2\right)^p \, dx \longrightarrow \int (d + e x)^{m+1} \text{PolynomialQuotient}[P_q[x], d + e x, x] \left(a + b x + c x^2\right)^p \, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+1)*PolynomialQuotient[Pq,d+e*x,x]*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,d+e*x,x],0]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+1)*PolynomialQuotient[Pq,d+e*x,x]*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,d+e*x,x],0]
```

2: $\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p \left(f+g\,x+h\,x^2\right) dx$

when $b\,e\,h\,(m+p+2)+2\,c\,d\,h\,(p+1)-c\,e\,g\,(m+2\,p+3)=0\,\land\,b\,d\,h\,(p+1)+a\,e\,h\,(m+1)-c\,e\,f\,(m+2\,p+3)=0\,\land\,m+2\,p+3\neq 0$

Derivation: Special case of one step of the Ostrogradskiy-Hermite integration method

Rule 1.2.1.9.2: If $b\,e\,h\,(m+p+2)+2\,c\,d\,h\,(p+1)-c\,e\,g\,(m+2\,p+3)=0\,\land$ $b\,d\,h\,(p+1)+a\,e\,h\,(m+1)-c\,e\,f\,(m+2\,p+3)=0\,\land\,m+2\,p+3\neq 0$, then

$$\int (d+e\,x)^m \left(a+b\,x+c\,x^2\right)^p \left(f+g\,x+h\,x^2\right) dx \,\longrightarrow\, \frac{h\,(d+e\,x)^{m+1}\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,e\,(m+2\,p+3)}$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*P2_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{f=Coeff[P2,x,0],g=Coeff[P2,x,1],h=Coeff[P2,x,2]},
  h*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(c*e*(m+2*p+3)) /;
 EqQ[b*e*h*(m+p+2)+2*c*d*h*(p+1)-c*e*g*(m+2*p+3),0] && EqQ[b*d*h*(p+1)+a*e*h*(m+1)-c*e*f*(m+2*p+3),0]] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[P2,x,2] && NeQ[m+2*p+3,0]
```

```
Int[(d_+e_.*x_)^m_.*P2_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{f=Coeff[P2,x,0],g=Coeff[P2,x,1],h=Coeff[P2,x,2]},
  h*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/(c*e*(m+2*p+3)) /;
 EqQ[2*d*h*(p+1)-e*g*(m+2*p+3),0] && EqQ[a*h*(m+1)-c*f*(m+2*p+3),0]] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[P2,x,2] && NeQ[m+2*p+3,0]
```

3: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $p+2\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.9.3: If $p+2\in\mathbb{Z}^+$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[(d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p,\,x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*Pq*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*Pq*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,m},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

4: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x \frac{\left(a+b\,x+c\,x^2\right)^p}{(b+2\,c\,x)^{2\,p}}==0$

Rule 1.2.1.9.4: If $b^2-4\,a\,c==0$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx \to \frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,(b+2\,c\,x)^{2\,\text{FracPart}[p]}}\int (d+e\,x)^m\,P_q[x]\,\left(b+2\,c\,x^2\right)^{2\,p}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x)^(2*FracPart[p]))*Int[(d+e*x)^m*Pq*(b+2*c*x)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[b^2-4*a*c,0]
```

5. $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0$

1: $\int (e\,x)^m\,P_q[x]\,\left(b\,x+c\,x^2\right)^p\,dx$ when $\text{PolynomialRemainder}[P_q[x],\,b+c\,x,\,x]==0$

Derivation: Algebraic simplification

Basis: $P_q[x]==\frac{1}{e\,x}\,\frac{e\,P_q[x]}{b+c\,x}\,\left(b\,x+c\,x^2\right)$

Rule 1.2.1.9.5.1: If $\text{PolynomialRemainder}[P_q[x],\,b+c\,x,\,x]==0$, then

$$\int (e\,x)^m\,P_q[x]\,\left(b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,e\int (e\,x)^{m-1}\,\text{PolynomialQuotient}[P_q[x],\,b+c\,x,\,x]\,\left(b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Pq_*(b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  e*Int[(e*x)^(m-1)*PolynomialQuotient[Pq,b+c*x,x]*(b*x+c*x^2)^(p+1),x] /;
FreeQ[{b,c,e,m,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,b+c*x,x],0]
```

2: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,\text{PolynomialRemainder}\left[P_q[x],\,a\,e+c\,d\,x,\,x\right]==0$

Derivation: Algebraic simplification

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $(d+e\,x)\,(a\,e+c\,d\,x)==d\,e\,\left(a+b\,x+c\,x^2\right)$

Rule 1.2.1.9.5.2: If

$b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,\text{PolynomialRemainder}\left[P_q[x],\,a\,e+c\,d\,x,\,x\right]==0$,

let $Q_{q-1}[x]\to\text{PolynomialQuotient}\left[P_q[x],\,a\,e+c\,d\,x,\,x\right]$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;d\,e\int (d+e\,x)^{m-1}\,Q_{q-1}[x]\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  d*e*Int[(d+e*x)^(m-1)*PolynomialQuotient[Pq,a*e+c*d*x,x]*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && EqQ[PolynomialRemainder[Pq,a*e+c*d*x,
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  d*e*Int[(d+e*x)^(m-1)*PolynomialQuotient[Pq,a*e+c*d*x,x]*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && EqQ[PolynomialRemainder[Pq,a*e+c*d*x,x],0]
```

3: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ c\,d^2-b\,d\,e+a\,e^2==0\ \wedge\ p+\frac{1}{2}\in\mathbb{Z}^-\ \wedge\ m>0$

Derivation: Algebraic expansion and special quadratic recurrence 2b

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $(d+e\,x)\,(a\,e+c\,d\,x)==d\,e\,\left(a+b\,x+c\,x^2\right)$

Rule 1.2.1.9.5.3: If $b^2-4\,a\,c\neq 0\ \wedge\ c\,d^2-b\,d\,e+a\,e^2==0\ \wedge\ p+\frac{1}{2}\in\mathbb{Z}^-\ \wedge\ m>0$,

let $Q_{q-1}[x]\to\text{PolynomialQuotient}[P_q[x],\,a\,e+c\,d\,x,\,x]$ and $f\to\text{PolynomialRemainder}[P_q[x],\,a\,e+c\,d\,x,\,x]$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\ \longrightarrow$$

$$f\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx+d\,e\int (d+e\,x)^{m-1}\,Q_{q-1}[x]\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx\ \longrightarrow$$

$$\frac{f\,(2\,c\,d-b\,e)\,(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}}{e\,(p+1)\,\left(b^2-4\,a\,c\right)}+$$

$$\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,\left(d\,e\,(p+1)\,\left(b^2-4\,a\,c\right)\,Q_{q-1}[x]-f\,(2\,c\,d-b\,e)\,(m+2\,p+2)\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a*e+c*d*x,x], f=PolynomialRemainder[Pq,a*e+c*d*x,x]},
  f*(2*c*d-b*e)*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(e*(p+1)*(b^2-4*a*c)) +
  1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)*
    ExpandToSum[d*e*(p+1)*(b^2-4*a*c)*Q-f*(2*c*d-b*e)*(m+2*p+2),x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[p+1/2,0] && GtQ[m,0]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a*e+c*d*x,x], f=PolynomialRemainder[Pq,a*e+c*d*x,x]},
  -d*f*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*a*e*(p+1)) +
  d/(2*a*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1)*ExpandToSum[2*a*e*(p+1)*Q+f*(m+2*p+2),x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && ILtQ[p+1/2,0] && GtQ[m,0]
```

4: $\int (d+e\,x)^m\, P_q[x]\, (a+b\,x+c\,x^2)^p\, dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ c\,d^2-b\,d\,e+a\,e^2==0\ \wedge\ m+q+2\,p+1==0\ \wedge\ m\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Rule 1.2.1.9.5.4: If $b^2-4\,a\,c\neq 0\ \wedge\ c\,d^2-b\,d\,e+a\,e^2==0\ \wedge\ m+q+2\,p+1==0\ \wedge\ m\in\mathbb{Z}^-$, then

$$\int (d+e\,x)^m\, P_q[x]\, (a+b\,x+c\,x^2)^p\, dx \longrightarrow \int (a+b\,x+c\,x^2)^p\, \text{ExpandIntegrand}\left[(d+e\,x)^m\, P_q[x],\, x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x+c*x^2)^p,(d+e*x)^m*Pq,x],x] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && EqQ[m+Expon[Pq,x]+2*p+1,0] && ILtQ[m,0]
```

```
Int[(d_+e_.*x_)^m_*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+c*x^2)^p,(d+e*x)^m*Pq,x],x] /;
FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && EqQ[m+Expon[Pq,x]+2*p+1,0] && ILtQ[m,0]
```

5: $\int (d+e\,x)^m\, P_q[x]\, (a+b\,x+c\,x^2)^p\, dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ c\,d^2-b\,d\,e+a\,e^2==0\ \wedge\ m+q+2\,p+1\neq 0$

Derivation: Algebraic expansion and quadratic recurrence 3a with $A=d$, $B=e$ and $m=m-1$

Rule 1.2.1.9.5.5: If $b^2-4\,a\,c\neq 0\ \wedge\ c\,d^2-b\,d\,e+a\,e^2==0\ \wedge\ m+q+2\,p+1\neq 0$, let $f\to P_q[x,\,q]$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\int (d+e\,x)^m\left(P_q[x]-\frac{f}{e^q}\,(d+e\,x)^q\right)\left(a+b\,x+c\,x^2\right)^p\,dx+\frac{f}{e^q}\int (d+e\,x)^{m+q}\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\frac{f\,(d+e\,x)^{m+q-1}\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,e^{q-1}\,(m+q+2\,p+1)}+\frac{1}{c\,e^q\,(m+q+2\,p+1)}\int (d+e\,x)^m\left(a+b\,x+c\,x^2\right)^p\cdot$$
$$\left(c\,e^q\,(m+q+2\,p+1)\,P_q[x]-c\,f\,(m+q+2\,p+1)\,(d+e\,x)^q+e\,f\,(m+p+q)\,(d+e\,x)^{q-2}\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\right)dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
  f*(d+e*x)^(m+q-1)*(a+b*x+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
  1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+b*x+c*x^2)^p*
    ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q+e*f*(m+p+q)*(d+e*x)^(q-2)*(b*d-2*a*e+(2*c*d-b*e)*x),x],x] /;
 NeQ[m+q+2*p+1,0]] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
  f*(d+e*x)^(m+q-1)*(a+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
  1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+c*x^2)^p*
    ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q-2*e*f*(m+p+q)*(d+e*x)^(q-2)*(a*e-c*d*x),x],x] /;
 NeQ[m+q+2*p+1,0]] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && Not[IGtQ[m,0]]
```

6: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $a+b\,x+c\,x^2==(d+e\,x)\left(\frac{a}{d}+\frac{c\,x}{e}\right)$

Rule 1.2.1.9.5.6: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int (d+e\,x)^{m+p}\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p P_q[x]\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,b,c,d,e,m},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,c,d,e,m},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

7: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $\partial_x\frac{\left(a+b\,x+c\,x^2\right)^p}{(d+e\,x)^p\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p}==0$

Rule 1.2.1.9.5.7: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\rightarrow\;\frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{(d+e\,x)^{\text{FracPart}[p]}\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^{\text{FracPart}[p]}}\int (d+e\,x)^{m+p}\,\left(\frac{a}{d}+\frac{c\,x}{e}\right)^p\,P_q[x]\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((d+e*x)^FracPart[p]*(a/d+(c*x)/e)^FracPart[p])*Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && Not[IGtQ[m,0]]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (a+c*x^2)^FracPart[p]/((d+e*x)^FracPart[p]*(a/d+(c*x)/e)^FracPart[p])*Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && Not[IGtQ[m,0]]
```

6. $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p<-1$

1: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p<-1\,\land\,m>0$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.2.1.9.6.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p<-1\,\land\,m>0$,
let $Q_{q-2}[x]\rightarrow\text{PolynomialQuotient}\left[P_q[x],\,a+b\,x+c\,x^2,\,x\right]$ and
$f+g\,x\rightarrow\text{PolynomialRemainder}\left[P_q[x],\,a+b\,x+c\,x^2,\,x\right]$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx+\int (d+e\,x)^{m-1}\,(d+e\,x)\,Q_{q-2}[x]\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx\;\longrightarrow$$

$$\frac{(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}\,\left(f\,b-2\,a\,g+(2\,c\,f-b\,g)\,x\right)}{(p+1)\,\left(b^2-4\,a\,c\right)}+$$
$$\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,\cdot$$

$$\left((p+1)\left(b^2-4ac\right)(d+ex)\,Q_{q-2}[x]+g\left(2aem+bd(2p+3)\right)-f\left(bem+2cd(2p+3)\right)-e\left(2cf-bg\right)(m+2p+3)\,x\right)dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x+c*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,1]},
  (d+e*x)^m*(a+b*x+c*x^2)^(p+1)*(f*b-2*a*g+(2*c*f-b*g)*x)/((p+1)*(b^2-4*a*c)) +
  1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)*
    ExpandToSum[(p+1)*(b^2-4*a*c)*(d+e*x)*Q+g*(2*a*e*m+b*d*(2*p+3))-f*(b*e*m+2*c*d*(2*p+3))-e*(2*c*f-b*g)*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && GtQ[m,0] &&
  (IntegerQ[p] || Not[IntegerQ[m]] || Not[RationalQ[a,b,c,d,e]]) &&
  Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+c*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,1]},
  (d+e*x)^m*(a+c*x^2)^(p+1)*(a*g-c*f*x)/(2*a*c*(p+1)) +
  1/(2*a*c*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1)*
    ExpandToSum[2*a*c*(p+1)*(d+e*x)*Q-a*e*g*m+c*d*f*(2*p+3)+c*e*f*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && GtQ[m,0] &&
  Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

2. $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p<-1\,\wedge\,m\not>0$

1: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p<-1\,\wedge\,m\in\mathbb{Z}^-$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.1.9.6.2.1: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p<-1\,\wedge\,m\in\mathbb{Z}^-$, let $Q_{m+q-2}[x]\to\text{PolynomialQuotient}\left[(d+e\,x)^m\,P_q[x],\,a+b\,x+c\,x^2,\,x\right]$ and $f+g\,x\to\text{PolynomialRemainder}\left[(d+e\,x)^m\,P_q[x],\,a+b\,x+c\,x^2,\,x\right]$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\int (f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx+\int Q_{m+q-2}[x]\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx\,\longrightarrow$$

$$\frac{\left(b\,f-2\,a\,g+\left(2\,c\,f-b\,g\right)\,x\right)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,\left(b^2-4\,a\,c\right)}+$$

$$\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}\,\left((p+1)\,\left(b^2-4\,a\,c\right)\,(d+e\,x)^{-m}\,Q_{m+q-2}[x]-(2\,p+3)\,\left(2\,c\,f-b\,g\right)\,(d+e\,x)^{-m}\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[(d+e*x)^m*Pq,a+b*x+c*x^2,x],
        f=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+b*x+c*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+b*x+c*x^2,x],x,1]},
  (b*f-2*a*g+(2*c*f-b*g)*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) +
  1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p+1)*
    ExpandToSum[(p+1)*(b^2-4*a*c)*(d+e*x)^(-m)*Q-(2*p+3)*(2*c*f-b*g)*(d+e*x)^(-m),x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && ILtQ[m,0]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[(d+e*x)^m*Pq,a+c*x^2,x],
        f=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+c*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+c*x^2,x],x,1]},
  (a*g-c*f*x)*(a+c*x^2)^(p+1)/(2*a*c*(p+1)) +
  1/(2*a*c*(p+1))*Int[(d+e*x)^m*(a+c*x^2)^(p+1)*
    ExpandToSum[2*a*c*(p+1)*(d+e*x)^(-m)*Q+c*f*(2*p+3)*(d+e*x)^(-m),x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && ILtQ[m,0]
```

2: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p<-1\,\wedge\,m\not>0$

Derivation: Algebraic expansion and quadratic recurrence 2b

Rule 1.2.1.9.6.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p<-1\,\wedge\,m\not>0$,

let $Q_{q-2}[x]\to \text{PolynomialQuotient}\left[P_q[x],\,a+b\,x+c\,x^2,\,x\right]$ and

$f+g\,x\to \text{PolynomialRemainder}\left[P_q[x],\,a+b\,x+c\,x^2,\,x\right]$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx+\int (d+e\,x)^m\,Q_{q-2}[x]\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx\,\longrightarrow$$

$$\Big(\Big((d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,\big(f\,\left(b\,c\,d-b^2\,e+2\,a\,c\,e\right)-a\,g\,\left(2\,c\,d-b\,e\right)+c\,\left(f\,\left(2\,c\,d-b\,e\right)-g\,\left(b\,d-2\,a\,e\right)\right)\,x\big)\Big)\Big/$$
$$\Big((p+1)\,\left(b^2-4\,a\,c\right)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)\Big)\Big)+$$
$$\frac{1}{(p+1)\,\left(b^2-4\,a\,c\right)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}\cdot$$
$$\Big((p+1)\,\left(b^2-4\,a\,c\right)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)\,Q_{q-2}[x]+$$
$$f\,\left(b\,c\,d\,e\,(2\,p-m+2)+b^2\,e^2\,(p+m+2)-2\,c^2\,d^2\,(2\,p+3)-2\,a\,c\,e^2\,(m+2\,p+3)\right)-$$
$$g\,\left(a\,e\,\left(b\,e-2\,c\,d\,m+b\,e\,m\right)-b\,d\,\left(3\,c\,d-b\,e+2\,c\,d\,p-b\,e\,p\right)\right)+$$
$$c\,e\,\left(g\,\left(b\,d-2\,a\,e\right)-f\,\left(2\,c\,d-b\,e\right)\right)\,(m+2\,p+4)\,x\Big)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x+c*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,1]},
   (d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)*(f*(b*c*d-b^2*e+2*a*c*e)-a*g*(2*c*d-b*e)+c*(f*(2*c*d-b*e)-g*(b*d-2*a*e))*x)/
     ((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)) +
  1/((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p+1)*
    ExpandToSum[(p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)*Q+
      f*(b*c*d*e*(2*p-m+2)+b^2*e^2*(p+m+2)-2*c^2*d^2*(2*p+3)-2*a*c*e^2*(m+2*p+3))-
      g*(a*e*(b*e-2*c*d*m+b*e*m)-b*d*(3*c*d-b*e+2*c*d*p-b*e*p))+
      c*e*(g*(b*d-2*a*e)-f*(2*c*d-b*e))*(m+2*p+4)*x,x],x]] /;
FreeQ[{a,b,c,d,e,m},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] &&
  Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+c*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,1]},
  -(d+e*x)^(m+1)*(a+c*x^2)^(p+1)*(a*(e*f-d*g)+(c*d*f+a*e*g)*x)/(2*a*(p+1)*(c*d^2+a*e^2)) +
  1/(2*a*(p+1)*(c*d^2+a*e^2))*Int[(d+e*x)^m*(a+c*x^2)^(p+1)*
    ExpandToSum[2*a*(p+1)*(c*d^2+a*e^2)*Q+c*d^2*f*(2*p+3)-a*e*(d*g*m-e*f*(m+2*p+3))+e*(c*d*f+a*e*g)*(m+2*p+4)*x,x],x]] /;
FreeQ[{a,c,d,e,m},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] &&
  Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

7: $\int (d+e x)^m P_q[x] \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m<-1$

Derivation: Algebraic expansion and quadratic recurrence 3b

Rule 1.2.1.9.7: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m<-1$,
let $Q_{q-1}[x] \to \text{PolynomialQuotient}[P_q[x], d+e x, x]$ and $R \to \text{PolynomialRemainder}[P_q[x], d+e x, x]$, then

$$\int (d+e x)^m P_q[x] \left(a+b x+c x^2\right)^p \, dx \longrightarrow$$

$$\int (d+e\,x)^{m+1}\,Q_{q-1}[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx+R\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$\frac{e\,R\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+$$

$$\frac{1}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int (d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p\,\cdot$$

$$\left((m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)\,Q_{q-1}[x]+c\,d\,R\,(m+1)-b\,e\,R\,(m+p+2)-c\,e\,R\,(m+2\,p+3)\,x\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,d+e*x,x], R=PolynomialRemainder[Pq,d+e*x,x]},
  (e*R*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1))/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  1/((m+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p*
     ExpandToSum[(m+1)*(c*d^2-b*d*e+a*e^2)*Q+c*d*R*(m+1)-b*e*R*(m+p+2)-c*e*R*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,b,c,d,e,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[m,-1]
```

```
Int[(d_+e_.*x_)^m_*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,d+e*x,x], R=PolynomialRemainder[Pq,d+e*x,x]},
  (e*R*(d+e*x)^(m+1)*(a+c*x^2)^(p+1))/((m+1)*(c*d^2+a*e^2)) +
  1/((m+1)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+1)*(a+c*x^2)^p*
     ExpandToSum[(m+1)*(c*d^2+a*e^2)*Q+c*d*R*(m+1)-c*e*R*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,c,d,e,p},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[m,-1]
```

8: $\int x^m P_q[x] \left(a+b x^2\right)^p \, dx$ when $\neg P_q\left[x^2\right] \land m+2 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $P_q[x] == \sum_{k=0}^{q/2} P_q[x, 2k] x^{2k} + x \sum_{k=0}^{(q-1)/2} P_q[x, 2k+1] x^{2k}$

Note: This rule transforms $x^m P_q[x]$ into a sum of the form $x^m Q_r\left[x^2\right] + x^{m+1} R_s\left[x^2\right]$.

Rule 1.2.1.9.8: If $\neg P_q\left[x^2\right] \land m+2 \in \mathbb{Z}^+$, then

$$\int x^m P_q[x] \left(a+b x^2\right)^p dx \to \int x^m \left(\sum_{k=0}^{\frac{q}{2}} P_q[x, 2k] x^{2k}\right) \left(a+b x^2\right)^p dx + \int x^{m+1} \left(\sum_{k=0}^{\frac{q-1}{2}} P_q[x, 2k+1] x^{2k}\right) \left(a+b x^2\right)^p dx$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],k},
  Int[x^m*Sum[Coeff[Pq,x,2*k]*x^(2*k),{k,0,q/2}]*(a+b*x^2)^p,x] +
  Int[x^(m+1)*Sum[Coeff[Pq,x,2*k+1]*x^(2*k),{k,0,(q-1)/2}]*(a+b*x^2)^p,x]] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && Not[PolyQ[Pq,x^2]] && IGtQ[m,-2] && Not[IntegerQ[2*p]]
```

9: $\int (d+e x)^m P_q[x] \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m+q+2 p+1 \neq 0$

Derivation: Algebraic expansion and quadratic recurrence 3a with $A = d$, $B = e$ and $m = m-1$

Rule 1.2.1.9.9: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m+q+2 p+1 \neq 0$, let $f \to P_q[x, q]$, then

$$\int (d+e x)^m P_q[x] \left(a+b x+c x^2\right)^p dx \to$$

$$\int (d+e x)^m \left(P_q[x] - \frac{f}{e^q} (d+e x)^q\right) \left(a+b x+c x^2\right)^p dx + \frac{f}{e^q} \int (d+e x)^{m+q} \left(a+b x+c x^2\right)^p dx \to$$

$$\frac{f\,(d+e\,x)^{m+q-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,e^{q-1}\,(m+q+2\,p+1)}+$$
$$\frac{1}{c\,e^q\,(m+q+2\,p+1)}\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,\left(c\,e^q\,(m+q+2\,p+1)\,P_q[x]-c\,f\,(m+q+2\,p+1)\,(d+e\,x)^q-\right.$$
$$\left.f\,(d+e\,x)^{q-2}\,\left(b\,d\,e\,(p+1)+a\,e^2\,(m+q-1)-c\,d^2\,(m+q+2\,p+1)-e\,(2\,c\,d-b\,e)\,(m+q+p)\,x\right)\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
  f*(d+e*x)^(m+q-1)*(a+b*x+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
  1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+b*x+c*x^2)^p*ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q-
    f*(d+e*x)^(q-2)*(b*d*e*(p+1)+a*e^2*(m+q-1)-c*d^2*(m+q+2*p+1)-e*(2*c*d-b*e)*(m+q+p)*x),x],x] /;
 GtQ[q,1] && NeQ[m+q+2*p+1,0]] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
  f*(d+e*x)^(m+q-1)*(a+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
  1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+c*x^2)^p*ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q-
    f*(d+e*x)^(q-2)*(a*e^2*(m+q-1)-c*d^2*(m+q+2*p+1)-2*c*d*e*(m+q+p)*x),x],x] /;
 GtQ[q,1] && NeQ[m+q+2*p+1,0]] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && Not[EqQ[d,0] && True] &&
  Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

10: $\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

Derivation: Algebraic expansion

Basis: $(d+e\,x)^m\,P_q[x] == \frac{P_q[x,q]\,(d+e\,x)^{m+q}}{e^q}+\frac{(d+e\,x)^m\,\left(e^q\,P_q[x]-P_q[x,q]\,(d+e\,x)^q\right)}{e^q}$

Rule 1.2.1.9.10: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0$, then

$$\int (d+e\,x)^m\,P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$
$$\frac{P_q[x,\,q]}{e^q}\int (d+e\,x)^{m+q}\,\left(a+b\,x+c\,x^2\right)^p\,dx+\frac{1}{e^q}\int (d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^p\,\left(e^q\,P_q[x]-P_q[x,\,q]\,(d+e\,x)^q\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Pq_*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  Coeff[Pq,x,q]/e^q*Int[(d+e*x)^(m+q)*(a+b*x+c*x^2)^p,x] +
  1/e^q*Int[(d+e*x)^m*(a+b*x+c*x^2)^p*ExpandToSum[e^q*Pq-Coeff[Pq,x,q]*(d+e*x)^q,x],x]] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  Coeff[Pq,x,q]/e^q*Int[(d+e*x)^(m+q)*(a+c*x^2)^p,x] +
  1/e^q*Int[(d+e*x)^m*(a+c*x^2)^p*ExpandToSum[e^q*Pq-Coeff[Pq,x,q]*(d+e*x)^q,x],x]] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] &&
  Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```