Rules for integrands of the form $P_q[x]\,(a+b\,x^2)^p$

1: $\int P_q[x]\,(a+b\,x^2)^p\,dx$ when $p\in\mathbb{Z}^+\land P_q[x,1]\neq 0$

Derivation: Algebraic expansion

Rule 1.1.2.x.1: If $p\in\mathbb{Z}^+\land P_q[x,1]\neq 0$, then

$$\int P_q[x]\,(a+b\,x^2)^p\,dx\;\longrightarrow\;\frac{P_q[x,1]\,(a+b\,x^2)^{p+1}}{2\,b\,(p+1)}+\int (P_q[x]-P_q[x,1]\,x)\,(a+b\,x^2)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  Coeff[Pq,x,1]*(a+b*x^2)^(p+1)/(2*b*(p+1)) +
  Int[ExpandToSum[Pq-Coeff[Pq,x,1]*x,x]*(a+b*x^2)^p,x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[p,0] && NeQ[Coeff[Pq,x,1],0]
```

2: $\int P_q[x]\,(a+b\,x^2)^p\,dx$ when $p+2\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.2.x.2: If $p+2\in\mathbb{Z}^+$, then

$$\int P_q[x]\,(a+b\,x^2)^p\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[P_q[x]\,(a+b\,x^2)^p,\,x\right]dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x^2)^p,x],x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

3: $\int P_q[x]\,\left(a+b\,x^2\right)^p\,dx$ when PolynomialRemainder[P_q[x], x, x] == 0

Derivation: Algebraic simplification

Rule 1.1.2.x.3: If PolynomialRemainder[P_q[x], x, x] == 0, then

$$\int P_q[x]\,\left(a+b\,x^2\right)^p\,dx \longrightarrow \int x\,\text{PolynomialQuotient}[P_q[x],\,x,\,x]\,\left(a+b\,x^2\right)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pq,x,x]*(a+b*x^2)^p,x] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,x,x],0] && Not[MatchQ[Pq,x^m_.*u_. /; IntegerQ[m]]]
```

4: $\int P_q[x^2] \left(a+b x^2\right)^p \, dx$ when $p+\frac{1}{2} \in \mathbb{Z}^- \land 2q+2p+1<0$

Derivation: Algebraic expansion and binomial recurrence 3b

Basis: $\int \left(a+b x^2\right)^p \, dx == \frac{x \left(a+b x^2\right)^{p+1}}{a} - \frac{b (2p+3)}{a} \int x^2 \left(a+b x^2\right)^p \, dx$

Note: Interestingly this rule eleminates the constant term of $P_q[x^2]$ rather than the highest degree term.

Rule 1.1.2.x.4: If $p+\frac{1}{2} \in \mathbb{Z}^- \land 2q+2p+1<0$, let $A \to P_q[x^2, 0]$ and $Q_{q-1}[x^2] \to \text{PolynomialQuotient}[P_q[x^2]-A, x^2, x]$, then

$$\int P_q[x^2] \left(a+b x^2\right)^p \, dx$$

$$\longrightarrow A \int \left(a+b x^2\right)^p \, dx + \int x^2 Q_{q-1}[x^2] \left(a+b x^2\right)^p \, dx$$

$$\longrightarrow \frac{A x \left(a+b x^2\right)^{p+1}}{a} + \frac{1}{a} \int x^2 \left(a+b x^2\right)^p \left(a Q_{q-1}[x^2] - A b (2p+3)\right) dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{A=Coeff[Pq,x,0],Q=PolynomialQuotient[Pq-Coeff[Pq,x,0],x^2,x]},
  A*x*(a+b*x^2)^(p+1)/a + 1/a*Int[x^2*(a+b*x^2)^p*(a*Q-A*b*(2*p+3)),x]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x^2] && ILtQ[p+1/2,0] && LtQ[Expon[Pq,x]+2*p+1,0]
```

5: $\int P_q[x] \left(a+b x^2\right)^p \, dx$ when $p<-1$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.1.2.x.5: If $p<-1$,
let $Q_{q-2}[x] \to \text{PolynomialQuotient}[P_q[x], a+b x^2, x]$ and $f+g x \to \text{PolynomialRemainder}[P_q[x], a+b x^2, x]$, then

$$\int P_q[x] \left(a+b\,x^2\right)^p dx \longrightarrow$$

$$\int (f+g\,x) \left(a+b\,x^2\right)^p dx + \int Q_{q-2}[x] \left(a+b\,x^2\right)^{p+1} dx \longrightarrow$$

$$\frac{(a\,g-b\,f\,x) \left(a+b\,x^2\right)^{p+1}}{2\,a\,b\,(p+1)} + \frac{1}{2\,a\,b\,(p+1)} \int \left(a+b\,x^2\right)^{p+1} \left(2\,a\,b\,(p+1)\,Q_{q-2}[x] + b\,f\,(2\,p+3)\right) dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x^2,x],
        f=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,0],
        g=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,1]},
  (a*g-b*f*x)*(a+b*x^2)^(p+1)/(2*a*b*(p+1)) +
  1/(2*a*b*(p+1))*Int[(a+b*x^2)^(p+1)*ExpandToSum[2*a*b*(p+1)*Q+b*f*(2*p+3),x],x]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && LtQ[p,-1]
```

6: $\int P_q[x] \left(a+b\,x^2\right)^p dx$ when $p \nless -1$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A = 0$, $B = 1$ and $m = m - n$

Reference: G&R 2.104

Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.

Rule 1.1.2.x.6: If $p \nless -1$, let $e \to P_q[x,\ q]$, then

$$\int P_q[x] \left(a+b\,x^2\right)^p dx \longrightarrow$$

$$\int \left(P_q[x] - e\, x^q\right) \left(a + b\, x^2\right)^p dx + e \int x^q \left(a + b\, x^2\right)^p dx \longrightarrow$$

$$\frac{e\, x^{q-1} \left(a + b\, x^2\right)^{p+1}}{b\, (q + 2\, p + 1)} + \frac{1}{b\, (q + 2\, p + 1)} \int \left(a + b\, x^2\right)^p \left(b\, (q + 2\, p + 1)\, P_q[x] - a\, e\, (q - 1)\, x^{q-2} - b\, e\, (q + 2\, p + 1)\, x^q\right) dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],e=Coeff[Pq,x,Expon[Pq,x]]},
  e*x^(q-1)*(a+b*x^2)^(p+1)/(b*(q+2*p+1)) +
  1/(b*(q+2*p+1))*Int[(a+b*x^2)^p*ExpandToSum[b*(q+2*p+1)*Pq-a*e*(q-1)*x^(q-2)-b*e*(q+2*p+1)*x^q,x],x]] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && Not[LeQ[p,-1]]
```