Rules for integrands of the form $(d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n$

1: $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])\,dx$

Derivation: Algebraic expansion and integration by substitution

Basis: $\text{Tan}[e+f\,x]\,F[\text{Sec}[e+f\,x]] == \frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,\text{Sec}[e+f\,x]\right]\,\partial_x\,\text{Sec}[e+f\,x]$

Rule:

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])\,dx \longrightarrow b\int \text{Tan}[e+f\,x]\,(d\,\text{Sec}[e+f\,x])^m\,dx + a\int (d\,\text{Sec}[e+f\,x])^m\,dx$$

$$\longrightarrow \frac{b\,(d\,\text{Sec}[e+f\,x])^m}{f\,m} + a\int (d\,\text{Sec}[e+f\,x])^m\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  b*(d*Sec[e+f*x])^m/(f*m) + a*Int[(d*Sec[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,m},x] && (IntegerQ[2*m] || NeQ[a^2+b^2,0])
```

2. $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0$

1: $\int \text{Sec}[e+f\,x]^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\,\land\,\frac{m}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $a^2+b^2==0\,\land\,\frac{m}{2}\in\mathbb{Z}$, then

$\text{Sec}[e+f\,x]^m\,(a+b\,\text{Tan}[e+f\,x])^n == \frac{1}{a^{m-2}\,b\,f}\,\text{Subst}\left[(a-x)^{m/2-1}\,(a+x)^{n+m/2-1},\,x,\,b\,\text{Tan}[e+f\,x]\right]\,\partial_x\,(b\,\text{Tan}[e+f\,x])$

Rule: If $a^2+b^2==0\,\land\,\frac{m}{2}\in\mathbb{Z}$, then

$$\int \text{Sec}[e+f\,x]^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{1}{a^{m-2}\,b\,f}\,\text{Subst}\left[\int (a-x)^{m/2-1}\,(a+x)^{n+m/2-1}\,dx,\,x,\,b\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  1/(a^(m-2)*b*f)*Subst[Int[(a-x)^(m/2-1)*(a+x)^(n+m/2-1),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,e,f,n},x] && EqQ[a^2+b^2,0] && IntegerQ[m/2]
```

2: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2==0 \land m+n==0$

Rule: If $a^2+b^2==0 \land m+n==0$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow \frac{b\,(d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n}{a\,f\,m}$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^n/(a*f*m) /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0] && EqQ[Simplify[m+n],0]
```

3. $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\,\land\,\frac{m}{2}+n==0$

1: $\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Tan}[e+f\,x]}}\,dx$ when $a^2+b^2==0$

Derivation: Integration by substitution

Basis: If $a^2+b^2==0$, then $\frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Tan}[e+f\,x]}}==-\frac{2\,a}{b\,f}\,\text{Subst}\left[\frac{1}{2-a\,x^2},\,x,\,\frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Tan}[e+f\,x]}}\right]\,\partial_x\,\frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Tan}[e+f\,x]}}$

Rule: If $a^2+b^2==0$, then

$$\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Tan}[e+f\,x]}}\,dx\,\longrightarrow\,-\frac{2\,a}{b\,f}\,\text{Subst}\left[\int\frac{1}{2-a\,x^2}\,dx,\,x,\,\frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Tan}[e+f\,x]}}\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]/Sqrt[a_+b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  -2*a/(b*f)*Subst[Int[1/(2-a*x^2),x],x,Sec[e+f*x]/Sqrt[a+b*Tan[e+f*x]]] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2+b^2,0]
```

2: $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\,\wedge\,\frac{m}{2}+n==0\,\wedge\,n>0$

Rule: If $a^2+b^2==0\,\wedge\,\frac{m}{2}+n==0\,\wedge\,n>0$, then

$$\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{b\,(d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n}{a\,f\,m}+\frac{a}{2\,d^2}\int (d\,\mathrm{Sec}[e+f\,x])^{m+2}\,(a+b\,\mathrm{Tan}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^n/(a*f*m) +
  a/(2*d^2)*Int[(d*Sec[e+f*x])^(m+2)*(a+b*Tan[e+f*x])^(n-1),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0] && EqQ[m/2+n,0] && GtQ[n,0]
```

3: $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\,\wedge\,\frac{m}{2}+n==0\,\wedge\,n<-1$

Rule: If $a^2+b^2==0\,\wedge\,\frac{m}{2}+n==0\,\wedge\,n<-1$, then

$$\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{2\,d^2\,(d\,\mathrm{Sec}[e+f\,x])^{m-2}\,(a+b\,\mathrm{Tan}[e+f\,x])^{n+1}}{b\,f\,(m-2)}+\frac{2\,d^2}{a}\int (d\,\mathrm{Sec}[e+f\,x])^{m-2}\,(a+b\,\mathrm{Tan}[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  2*d^2*(d*Sec[e+f*x])^(m-2)*(a+b*Tan[e+f*x])^(n+1)/(b*f*(m-2)) +
  2*d^2/a*Int[(d*Sec[e+f*x])^(m-2)*(a+b*Tan[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0] && EqQ[m/2+n,0] && LtQ[n,-1]
```

4: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2=0 \land \frac{m}{2}+n=0$

Derivation: Piecewise constant extraction

Basis: If $a^2+b^2=0$, then $\partial_x \frac{(a+b \operatorname{Tan}[e+f x])^n (a-b \operatorname{Tan}[e+f x])^n}{(d \operatorname{Sec}[e+f x])^{2n}}=0$

Note: Degree of secant factor in resulting integrand is even, making it easy to integrate by substitution.

Rule: If $a^2+b^2=0 \land \frac{m}{2}+n=0$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow$$

$$\left(\left(\left(\frac{a}{d}\right)^{2 \operatorname{IntPart}[n]} (a+b \operatorname{Tan}[e+f x])^{\operatorname{FracPart}[n]} (a-b \operatorname{Tan}[e+f x])^{\operatorname{FracPart}[n]}\right) \Big/ (d \operatorname{Sec}[e+f x])^{2 \operatorname{FracPart}[n]}\right) \int \frac{1}{(a-b \operatorname{Tan}[e+f x])^n} \, dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a/d)^(2*IntPart[n])*(a+b*Tan[e+f*x])^FracPart[n]*(a-b*Tan[e+f*x])^FracPart[n]/(d*Sec[e+f*x])^(2*FracPart[n])*
    Int[1/(a-b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0] && EqQ[Simplify[m/2+n],0]
```

4. $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\ \land\ \frac{m}{2}+n\in\mathbb{Z}^+$

1: $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\ \land\ \frac{m}{2}+n==1$

Rule: If $a^2+b^2==0\ \land\ \frac{m}{2}+n==1$, then

$$\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx \longrightarrow \frac{2\,b\,(d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^{n-1}}{f\,m}$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  2*b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1)/(f*m) /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0] && EqQ[Simplify[m/2+n-1],0]
```

2: $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\ \land\ \frac{m}{2}+n-1\in\mathbb{Z}^+\ \land\ n\notin\mathbb{Z}$

Rule: If $a^2+b^2==0\ \land\ \frac{m}{2}+n-1\in\mathbb{Z}^+\ \land\ n\notin\mathbb{Z}$, then

$$\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx \longrightarrow$$
$$\frac{b\,(d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^{n-1}}{f\,(m+n-1)}+\frac{a\,(m+2\,n-2)}{m+n-1}\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1)/(f*(m+n-1)) +
  a*(m+2*n-2)/(m+n-1)*Int[(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0] && IGtQ[Simplify[m/2+n-1],0] && Not[IntegerQ[n]]
```

5. $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\land n>0$

1: $\int \sqrt{d\,\text{Sec}[e+f\,x]}\,\sqrt{a+b\,\text{Tan}[e+f\,x]}\,dx$ when $a^2+b^2==0$

Derivation: Integration by substitution

Basis: If $a^2+b^2==0$, then $\sqrt{d\,\text{Sec}[e+f\,x]}\,\sqrt{a+b\,\text{Tan}[e+f\,x]}==-\frac{4\,b\,d^2}{f}\,\text{Subst}\left[\frac{x^2}{a^2+d^2\,x^4},\,x,\,\frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}\right]\partial_x\frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}$

Rule: If $a^2+b^2==0$, then

$$\int\sqrt{d\,\text{Sec}[e+f\,x]}\,\sqrt{a+b\,\text{Tan}[e+f\,x]}\,dx\to-\frac{4\,b\,d^2}{f}\,\text{Subst}\left[\int\frac{x^2}{a^2+d^2\,x^4}\,dx,\,x,\,\frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{d\,\text{Sec}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[d_.*sec[e_.+f_.*x_]]*Sqrt[a_+b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  -4*b*d^2/f*Subst[Int[x^2/(a^2+d^2*x^4),x],x,Sqrt[a+b*Tan[e+f*x]]/Sqrt[d*Sec[e+f*x]]] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0]
```

2: $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\,\wedge\,n>1\,\wedge\,m<0$

Rule: If $a^2+b^2==0\,\wedge\,n>1\,\wedge\,m<0$, then

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{2\,b\,(d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^{n-1}}{f\,m}-\frac{b^2\,(m+2\,n-2)}{d^2\,m}\int (d\,\text{Sec}[e+f\,x])^{m+2}\,(a+b\,\text{Tan}[e+f\,x])^{n-2}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  2*b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1)/(f*m) -
  b^2*(m+2*n-2)/(d^2*m)*Int[(d*Sec[e+f*x])^(m+2)*(a+b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0] && GtQ[n,1] && (IGtQ[n/2,0] && ILtQ[m-1/2,0] || EqQ[n,2] && LtQ[m,0] ||
  LeQ[m,-1] && GtQ[m+n,0] || ILtQ[m,0] && LtQ[m/2+n-1,0] && IntegerQ[n] || EqQ[n,3/2] && EqQ[m,-1/2]) && IntegerQ[2*m]
```

3: $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\wedge n>0\wedge m<-1$

Rule: If $a^2+b^2==0\wedge n>0\wedge m<-1$, then

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx\,\rightarrow$$
$$\frac{b\,(d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n}{a\,f\,m}+\frac{a\,(m+n)}{m\,d^2}\int (d\,\text{Sec}[e+f\,x])^{m+2}\,(a+b\,\text{Tan}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^n/(a*f*m) +
  a*(m+n)/(m*d^2)*Int[(d*Sec[e+f*x])^(m+2)*(a+b*Tan[e+f*x])^(n-1),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0] && GtQ[n,0] && LtQ[m,-1] && IntegersQ[2*m,2*n]
```

4: $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\wedge n>0$

Rule: If $a^2+b^2==0\wedge n>0$, then

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx\,\rightarrow$$
$$\frac{b\,(d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^{n-1}}{f\,(m+n-1)}+\frac{a\,(m+2\,n-2)}{m+n-1}\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1)/(f*(m+n-1)) +
  a*(m+2*n-2)/(m+n-1)*Int[(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1),x] /;
FreeQ[{a,b,d,e,f,m},x] && EqQ[a^2+b^2,0] && GtQ[n,0] && NeQ[m+n-1,0] && IntegersQ[2*m,2*n]
```

6. $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2==0 \land n<0$

1: $\int \frac{(d \operatorname{Sec}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Tan}[e+f x]}} \, dx$ when $a^2+b^2==0$

Derivation: Piecewise constant extraction

Basis: If $a^2+b^2==0$, then $\partial_x \frac{\operatorname{Sec}[e+f x]}{\sqrt{a-b \operatorname{Tan}[e+f x]} \sqrt{a+b \operatorname{Tan}[e+f x]}}==0$

Rule: If $a^2+b^2==0$, then

$$\int \frac{(d \operatorname{Sec}[e+f x])^{3/2}}{\sqrt{a+b \operatorname{Tan}[e+f x]}} \, dx \to \frac{d \operatorname{Sec}[e+f x]}{\sqrt{a-b \operatorname{Tan}[e+f x]} \sqrt{a+b \operatorname{Tan}[e+f x]}} \int \sqrt{d \operatorname{Sec}[e+f x]} \sqrt{a-b \operatorname{Tan}[e+f x]} \, dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^(3/2)/Sqrt[a_+b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  d*Sec[e+f*x]/(Sqrt[a-b*Tan[e+f*x]]*Sqrt[a+b*Tan[e+f*x]])*Int[Sqrt[d*Sec[e+f*x]]*Sqrt[a-b*Tan[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0]
```

2: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2==0 \land n<-1 \land m>1$

Rule: If $a^2+b^2==0 \land n<-1 \land m>1$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow$$
$$\frac{2 d^2 (d \operatorname{Sec}[e+f x])^{m-2} (a+b \operatorname{Tan}[e+f x])^{n+1}}{b f (m+2 n)}-\frac{d^2 (m-2)}{b^2 (m+2 n)} \int (d \operatorname{Sec}[e+f x])^{m-2} (a+b \operatorname{Tan}[e+f x])^{n+2} \, dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  2*d^2*(d*Sec[e+f*x])^(m-2)*(a+b*Tan[e+f*x])^(n+1)/(b*f*(m+2*n)) -
  d^2*(m-2)/(b^2*(m+2*n))*Int[(d*Sec[e+f*x])^(m-2)*(a+b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,d,e,f,m},x] && EqQ[a^2+b^2,0] && LtQ[n,-1] &&
  (ILtQ[n/2,0] && IGtQ[m-1/2,0] || EqQ[n,-2] || IGtQ[m+n,0] || IntegersQ[n,m+1/2] && GtQ[2*m+n+1,0]) && IntegerQ[2*m]
```

3: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2=0 \land n<0 \land m>1$

Rule: If $a^2+b^2=0 \land n<0 \land m>1$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx \to$$
$$\frac{d^2 (d \operatorname{Sec}[e+f x])^{m-2} (a+b \operatorname{Tan}[e+f x])^{n+1}}{b f (m+n-1)}+\frac{d^2 (m-2)}{a (m+n-1)} \int (d \operatorname{Sec}[e+f x])^{m-2} (a+b \operatorname{Tan}[e+f x])^{n+1} \, dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  d^2*(d*Sec[e+f*x])^(m-2)*(a+b*Tan[e+f*x])^(n+1)/(b*f*(m+n-1)) +
  d^2*(m-2)/(a*(m+n-1))*Int[(d*Sec[e+f*x])^(m-2)*(a+b*Tan[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0] && LtQ[n,0] && GtQ[m,1] && Not[ILtQ[m+n,0]] && NeQ[m+n-1,0] && IntegersQ[2*m,2*n]
```

4: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2=0 \land n<0$

Rule: If $a^2+b^2=0 \land n<0$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx \to$$
$$\frac{a (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n}{b f (m+2 n)}+\frac{m+n}{a (m+2 n)} \int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^{n+1} \, dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^n/(b*f*(m+2*n)) +
   Simplify[m+n]/(a*(m+2*n))*Int[(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,m},x] && EqQ[a^2+b^2,0] && LtQ[n,0] && NeQ[m+2*n,0] && IntegersQ[2*m,2*n]
```

7. $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0 \land m+n\in\mathbb{Z}$

1: $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0 \land m+n-1\in\mathbb{Z}^+$

Rule: If $a^2+b^2==0 \land m+n-1\in\mathbb{Z}^+$, then

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx \longrightarrow$$
$$\frac{b\,(d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^{n-1}}{f\,(m+n-1)}+\frac{a\,(m+2\,n-2)}{m+n-1}\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1)/(f*Simplify[m+n-1]) +
  a*(m+2*n-2)/Simplify[m+n-1]*Int[(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n-1),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0] && IGtQ[Simplify[m+n-1],0] && RationalQ[n]
```

2: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2==0 \land m+n \in \mathbb{Z}^-$

Rule: If $a^2+b^2==0 \land m+n \in \mathbb{Z}^-$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx \to$$
$$\frac{a (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n}{b f (m+2 n)}+\frac{m+n}{a (m+2 n)} \int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^{n+1} \, dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^n/(b*f*(m+2*n)) +
  Simplify[m+n]/(a*(m+2*n))*Int[(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0] && ILtQ[Simplify[m+n],0] && NeQ[m+2*n,0]
```

x: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2==0 \land n \in \mathbb{Z}$

Derivation: Piecewise constant extraction, algebraic expansion and integration by substitution

Basis: $\partial_x \frac{(d \operatorname{Sec}[e+f x])^m}{\left(\operatorname{Sec}[e+f x]^2\right)^{m/2}}==0$

Basis: $\operatorname{Sec}[e+f x]^2==1+\operatorname{Tan}[e+f x]^2$

Basis: $F[b \operatorname{Tan}[e+f x]]==\frac{1}{b f} \operatorname{Subst}\left[\frac{F[x]}{1+\frac{x^2}{b^2}}, x, b \operatorname{Tan}[e+f x]\right] \partial_x (b \operatorname{Tan}[e+f x])$

Rule: If $a^2+b^2==0 \land n \in \mathbb{Z}$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n \, dx \to \frac{(d \operatorname{Sec}[e+f x])^m}{\left(\operatorname{Sec}[e+f x]^2\right)^{m/2}} \int (a+b \operatorname{Tan}[e+f x])^n \left(1+\operatorname{Tan}[e+f x]^2\right)^{m/2} \, dx$$

$$\longrightarrow \frac{(d\,\text{Sec}[e+f\,x])^m}{b\,f\,\left(\text{Sec}[e+f\,x]^2\right)^{m/2}}\,\text{Subst}\left[\int (a+x)^n\left(1-\frac{x^2}{a^2}\right)^{m/2-1}dx,\,x,\,b\,\text{Tan}[e+f\,x]\right]$$

$$\longrightarrow \frac{a^n\,(d\,\text{Sec}[e+f\,x])^m}{b\,f\,\left(\text{Sec}[e+f\,x]^2\right)^{m/2}}\,\text{Subst}\left[\int\left(1+\frac{x}{a}\right)^{n+m/2-1}\left(1-\frac{x}{a}\right)^{m/2-1}dx,\,x,\,b\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
(* Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a^n*(d*Sec[e+f*x])^m/(b*f*(Sec[e+f*x]^2)^(m/2))*Subst[Int[(1+x/a)^(n+m/2-1)*(1-x/a)^(m/2-1),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,d,e,f,m},x] && EqQ[a^2+b^2,0] && IntegerQ[n] *)
```

x: $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0$

Derivation: Piecewise constant extraction, algebraic expansion and integration by substitution

Basis: $\partial_x \frac{(d\,\mathrm{Sec}[e+f\,x])^m}{(\mathrm{Sec}[e+f\,x]^2)^{m/2}}==0$

Basis: $\mathrm{Sec}[e+f\,x]^2==1+\mathrm{Tan}[e+f\,x]^2$

Basis: $F[b\,\mathrm{Tan}[e+f\,x]]==\frac{1}{b\,f}\,\mathrm{Subst}\left[\frac{F[x]}{1+\frac{x^2}{b^2}},\,x,\,b\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\,(b\,\mathrm{Tan}[e+f\,x])$

Rule: If $a^2+b^2==0$, then

$$\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{(d\,\mathrm{Sec}[e+f\,x])^m}{(\mathrm{Sec}[e+f\,x]^2)^{m/2}}\int (a+b\,\mathrm{Tan}[e+f\,x])^n\,(1+\mathrm{Tan}[e+f\,x]^2)^{m/2}\,dx$$

$$\longrightarrow\,\frac{(d\,\mathrm{Sec}[e+f\,x])^m}{b\,f\,(\mathrm{Sec}[e+f\,x]^2)^{m/2}}\,\mathrm{Subst}\left[\int (a+x)^n\left(1+\frac{x^2}{b^2}\right)^{m/2-1}dx,\,x,\,b\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
(* Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (d*Sec[e+f*x])^m/(b*f*(Sec[e+f*x]^2)^(m/2))*Subst[Int[(a+x)^n*(1+x^2/b^2)^(m/2-1),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0] *)
```

8: $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0$

Derivation: Piecewise constant extraction

Basis: If $a^2+b^2==0$, then $\partial_x \frac{(d\,\text{Sec}[e+f\,x])^m}{(a+b\,\text{Tan}[e+f\,x])^{m/2}\,(a-b\,\text{Tan}[e+f\,x])^{m/2}}==0$

Rule: If $a^2+b^2==0$, then

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx \to$$
$$\frac{(d\,\text{Sec}[e+f\,x])^m}{(a+b\,\text{Tan}[e+f\,x])^{m/2}\,(a-b\,\text{Tan}[e+f\,x])^{m/2}}\int (a+b\,\text{Tan}[e+f\,x])^{m/2+n}\,(a-b\,\text{Tan}[e+f\,x])^{m/2}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  (d*Sec[e+f*x])^m/((a+b*Tan[e+f*x])^(m/2)*(a-b*Tan[e+f*x])^(m/2))*Int[(a+b*Tan[e+f*x])^(m/2+n)*(a-b*Tan[e+f*x])^(m/2),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2+b^2,0]
```

3. $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2\neq 0$

1: $\int \text{Sec}[e+f\,x]^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2\neq 0 \land \frac{m}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion and integration by substitution

Basis: $\text{Sec}[e+f\,x]^2==1+\text{Tan}[e+f\,x]^2$

Basis: $F[b\,\text{Tan}[e+f\,x]]==\frac{1}{b\,f}\,\text{Subst}\left[\frac{F[x]}{1+\frac{x^2}{b^2}},\,x,\,b\,\text{Tan}[e+f\,x]\right]\partial_x(b\,\text{Tan}[e+f\,x])$

Rule: If $a^2+b^2\neq 0 \land \frac{m}{2}\in\mathbb{Z}$, then

$$\int \text{Sec}[e+f x]^m (a+b \text{Tan}[e+f x])^n \, dx \to \int (a+b \text{Tan}[e+f x])^n \left(1+\text{Tan}[e+f x]^2\right)^{m/2} dx$$

$$\to \frac{1}{b f} \text{Subst}\left[\int (a+x)^n \left(1+\frac{x^2}{b^2}\right)^{\frac{m}{2}-1} dx,\ x,\ b \text{Tan}[e+f x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  1/(b*f)*Subst[Int[(a+x)^n*(1+x^2/b^2)^(m/2-1),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,e,f,n},x] && NeQ[a^2+b^2,0] && IntegerQ[m/2]
```

2. $\int (d \text{Sec}[e+f x])^m (a+b \text{Tan}[e+f x])^2 \, dx$ when $a^2+b^2 \neq 0$

1: $\int \frac{(a+b \text{Tan}[e+f x])^2}{\text{Sec}[e+f x]} dx$ when $a^2+b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{(a+b \text{Tan}[e+f x])^2}{\text{Sec}[e+f x]} == b^2 \text{Sec}[e+f x] + 2 a b \text{Sin}[e+f x] + \left(a^2-b^2\right) \text{Cos}[e+f x]$

Rule: If $a^2+b^2 \neq 0$, then

$$\int \frac{(a+b \text{Tan}[e+f x])^2}{\text{Sec}[e+f x]} dx \to b^2 \int \text{Sec}[e+f x] \, dx + 2 a b \int \text{Sin}[e+f x] \, dx + \left(a^2-b^2\right) \int \text{Cos}[e+f x] \, dx$$

$$\to \frac{b^2 \text{ArcTanh}[\text{Sin}[e+f x]]}{f} - \frac{2 a b \text{Cos}[e+f x]}{f} + \frac{\left(a^2-b^2\right) \text{Sin}[e+f x]}{f}$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^2/sec[e_.+f_.*x_],x_Symbol] :=
  b^2*ArcTanh[Sin[e+f*x]]/f - 2*a*b*Cos[e+f*x]/f + (a^2-b^2)*Sin[e+f*x]/f /;
FreeQ[{a,b,e,f},x] && NeQ[a^2+b^2,0]
```

2: $\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^2 \, dx$ when $a^2+b^2 \neq 0 \land m \neq -1$

Rule: If $a^2+b^2 \neq 0 \land m \neq -1$, then

$$\int (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^2 \, dx \to$$

$$\frac{b (d \operatorname{Sec}[e+f x])^m (a+b \operatorname{Tan}[e+f x])}{f (m+1)}+\frac{1}{m+1} \int (d \operatorname{Sec}[e+f x])^m \left(a^2 (m+1)-b^2+a b (m+2) \operatorname{Tan}[e+f x]\right) dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^2,x_Symbol] :=
  b*(d*Sec[e+f*x])^m*(a+b*Tan[e+f*x])/(f*(m+1)) +
  1/(m+1)*Int[(d*Sec[e+f*x])^m*(a^2*(m+1)-b^2+a*b*(m+2)*Tan[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,m},x] && NeQ[a^2+b^2,0] && NeQ[m,-1]
```

3. $\int \frac{(d \operatorname{Sec}[e+f x])^m}{a+b \operatorname{Tan}[e+f x]} \, dx$ when $a^2+b^2 \neq 0 \land m \in \mathbb{Z}$

1. $\int \frac{(d \operatorname{Sec}[e+f x])^m}{a+b \operatorname{Tan}[e+f x]} \, dx$ when $a^2+b^2 \neq 0 \land m \in \mathbb{Z}^+$

1: $\int \frac{\operatorname{Sec}[e+f x]}{a+b \operatorname{Tan}[e+f x]} \, dx$ when $a^2+b^2 \neq 0$

Derivation: Integration by substitution

Basis: $\frac{\operatorname{Sec}[e+f x]}{a+b \operatorname{Tan}[e+f x]} = -\frac{1}{f} \operatorname{Subst}\left[\frac{1}{a^2+b^2-x^2}, x, \frac{b-a \operatorname{Tan}[e+f x]}{\operatorname{Sec}[e+f x]}\right] \partial_x \frac{b-a \operatorname{Tan}[e+f x]}{\operatorname{Sec}[e+f x]}$

Rule: If $a^2+b^2 \neq 0$, then

$$\int \frac{\text{Sec}[e+f\,x]}{a+b\,\text{Tan}[e+f\,x]}\,dx \longrightarrow -\frac{1}{f}\,\text{Subst}\left[\int \frac{1}{a^2+b^2-x^2}\,dx,\,x,\,\frac{b-a\,\text{Tan}[e+f\,x]}{\text{Sec}[e+f\,x]}\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -1/f*Subst[Int[1/(a^2+b^2-x^2),x],x,(b-a*Tan[e+f*x])/Sec[e+f*x]] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2+b^2,0]
```

2: $\int \frac{(d\,\text{Sec}[e+f\,x])^m}{a+b\,\text{Tan}[e+f\,x]}\,dx$ when $a^2+b^2\neq 0 \land m-1\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\text{Sec}[c+x]^2}{a+b\,\text{Tan}[c+x]} == -\frac{a-b\,\text{Tan}[c+x]}{b^2}+\frac{a^2+b^2}{b^2\,(a+b\,\text{Tan}[c+x])}$

Rule: If $a^2+b^2\neq 0 \land m-1\in\mathbb{Z}^+$, then

$$\int \frac{(d\,\text{Sec}[e+f\,x])^m}{a+b\,\text{Tan}[e+f\,x]}\,dx \longrightarrow -\frac{d^2}{b^2}\int (d\,\text{Sec}[e+f\,x])^{m-2}\,(a-b\,\text{Tan}[e+f\,x])\,dx+\frac{d^2\,(a^2+b^2)}{b^2}\int \frac{(d\,\text{Sec}[e+f\,x])^{m-2}}{a+b\,\text{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -d^2/b^2*Int[(d*Sec[e+f*x])^(m-2)*(a-b*Tan[e+f*x]),x] +
  d^2*(a^2+b^2)/b^2*Int[(d*Sec[e+f*x])^(m-2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2+b^2,0] && IGtQ[m,1]
```

2: $\int \frac{(d \operatorname{Sec}[e+f x])^m}{a+b \operatorname{Tan}[e+f x]} \, dx$ when $a^2+b^2 \neq 0 \land m \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b \operatorname{Tan}[e+f x]} == \frac{a-b \operatorname{Tan}[e+f x]}{a^2+b^2} + \frac{b^2 \operatorname{Sec}[e+f x]^2}{(a^2+b^2)\,(a+b \operatorname{Tan}[e+f x])}$

Rule: If $a^2+b^2 \neq 0 \land m \in \mathbb{Z}^-$, then

$$\int \frac{(d \operatorname{Sec}[e+f x])^m}{a+b \operatorname{Tan}[e+f x]} \, dx \rightarrow \frac{1}{a^2+b^2} \int (d \operatorname{Sec}[e+f x])^m \,(a-b \operatorname{Tan}[e+f x]) \, dx + \frac{b^2}{d^2\,(a^2+b^2)} \int \frac{(d \operatorname{Sec}[e+f x])^{m+2}}{a+b \operatorname{Tan}[e+f x]} \, dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(a^2+b^2)*Int[(d*Sec[e+f*x])^m*(a-b*Tan[e+f*x]),x] +
  b^2/(d^2*(a^2+b^2))*Int[(d*Sec[e+f*x])^(m+2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2+b^2,0] && ILtQ[m,0]
```

4: $\int (d \operatorname{Sec}[e+f x])^m \,(a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2 \neq 0 \land \frac{m}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction, algebraic expansion and integration by substitution

Basis: $\partial_x \frac{(d \operatorname{Sec}[e+f x])^m}{(\operatorname{Sec}[e+f x]^2)^{m/2}} == 0$

Basis: $\operatorname{Sec}[e+f x]^2 == 1+\operatorname{Tan}[e+f x]^2$

Basis: $F[b \operatorname{Tan}[e+f x]] == \frac{1}{b f} \operatorname{Subst}\left[\frac{F[x]}{1+\frac{x^2}{b^2}}, x, b \operatorname{Tan}[e+f x]\right] \partial_x (b \operatorname{Tan}[e+f x])$

Rule: If $a^2+b^2 \neq 0 \land \frac{m}{2} \notin \mathbb{Z}$, then

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx \longrightarrow \frac{d^{2\,\text{IntPart}[m/2]}\,(d\,\text{Sec}[e+f\,x])^{2\,\text{FracPart}[m/2]}}{\left(\text{Sec}[e+f\,x]^2\right)^{\text{FracPart}[m/2]}}\int (a+b\,\text{Tan}[e+f\,x])^n\,\left(1+\text{Tan}[e+f\,x]^2\right)^{m/2}\,dx$$

$$\longrightarrow \frac{d^{2\,\text{IntPart}[m/2]}\,(d\,\text{Sec}[e+f\,x])^{2\,\text{FracPart}[m/2]}}{b\,f\,\left(\text{Sec}[e+f\,x]^2\right)^{\text{FracPart}[m/2]}}\,\text{Subst}\left[\int (a+x)^n\,\left(1+\frac{x^2}{b^2}\right)^{\frac{m}{2}-1}\,dx,\,x,\,b\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  d^(2*IntPart[m/2])*(d*Sec[e+f*x])^(2*FracPart[m/2])/(b*f*(Sec[e+f*x]^2)^FracPart[m/2])*
    Subst[Int[(a+x)^n*(1+x^2/b^2)^(m/2-1),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n},x] && NeQ[a^2+b^2,0] && Not[IntegerQ[m/2]]
```

Rules for integrands of the form $(d\,\text{Cos}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n$

1. $\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $m\notin\mathbb{Z}$

1: $\int \frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{d\,\text{Cos}[e+f\,x]}}\,dx$ when $a^2+b^2==0$

Derivation: Integration by substitution

Basis: If $a^2+b^2==0$, then
$\frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{d\,\text{Cos}[e+f\,x]}}==-\frac{4\,b}{f}\,\text{Subst}\left[\frac{x^2}{a^2\,d^2+x^4},\,x,\,\sqrt{d\,\text{Cos}[e+f\,x]}\,\sqrt{a+b\,\text{Tan}[e+f\,x]}\right]\,\partial_x\left(\sqrt{d\,\text{Cos}[e+f\,x]}\,\sqrt{a+b\,\text{Tan}[e+f\,x]}\right)$

Rule: If $a^2+b^2==0$, then

$$\int \frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{d\,\text{Cos}[e+f\,x]}}\,dx\;\longrightarrow\;-\frac{4\,b}{f}\,\text{Subst}\left[\int\frac{x^2}{a^2\,d^2+x^4}\,dx,\,x,\,\sqrt{d\,\text{Cos}[e+f\,x]}\,\sqrt{a+b\,\text{Tan}[e+f\,x]}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*tan[e_.+f_.*x_]]/Sqrt[d_.cos[e_.+f_.*x_]],x_Symbol] :=
  -4*b/f*Subst[Int[x^2/(a^2*d^2+x^4),x],x,Sqrt[d*Cos[e+f*x]]*Sqrt[a+b*Tan[e+f*x]]] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0]
```

2: $\int \frac{1}{(d\,\mathrm{Cos}[e+f\,x])^{3/2}\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx$ when $a^2+b^2==0$

Derivation: Piecewise constant extraction

Basis: If $a^2+b^2==0$, then $\partial_x \frac{1}{\mathrm{Cos}[e+f\,x]\sqrt{a-b\,\mathrm{Tan}[e+f\,x]}\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}==0$

Rule: If $a^2+b^2==0$, then

$$\int \frac{1}{(d\,\mathrm{Cos}[e+f\,x])^{3/2}\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx \longrightarrow \frac{1}{d\,\mathrm{Cos}[e+f\,x]\sqrt{a-b\,\mathrm{Tan}[e+f\,x]}\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\int \frac{\sqrt{a-b\,\mathrm{Tan}[e+f\,x]}}{\sqrt{d\,\mathrm{Cos}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/((d_.*cos[e_.+f_.*x_])^(3/2)*Sqrt[a_+b_.*tan[e_.+f_.*x_]]),x_Symbol] :=
  1/(d*Cos[e+f*x]*Sqrt[a-b*Tan[e+f*x]]*Sqrt[a+b*Tan[e+f*x]])*Int[Sqrt[a-b*Tan[e+f*x]]/Sqrt[d*Cos[e+f*x]],x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2+b^2,0]
```

3: $\int (d \operatorname{Cos}[e + f x])^m (a + b \operatorname{Tan}[e + f x])^n \, dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((d \operatorname{Cos}[e + f x])^m (d \operatorname{Sec}[e + f x])^m\right) == 0$

Rule: If $m \notin \mathbb{Z}$, then

$$\int (d \operatorname{Cos}[e + f x])^m (a + b \operatorname{Tan}[e + f x])^n \, dx \longrightarrow (d \operatorname{Cos}[e + f x])^m (d \operatorname{Sec}[e + f x])^m \int \frac{(a + b \operatorname{Tan}[e + f x])^n}{(d \operatorname{Sec}[e + f x])^m} \, dx$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  (d*Cos[e+f*x])^m*(d*Sec[e+f*x])^m*Int[(a+b*Tan[e+f*x])^n/(d*Sec[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,m,n},x] && Not[IntegerQ[m]]
```