

Rules for integrands of the form $(d + e x)^m P_q[x] (a + b x + c x^2)^p$ when $q > 1$

1: $\int (d + e x)^m P_q[x] (a + b x + c x^2)^p dx$ when $\text{PolynomialRemainder}[P_q[x], d + e x, x] == 0$

Derivation: Algebraic simplification

Rule 1.2.1.9.1: If $\text{PolynomialRemainder}[P_q[x], d + e x, x] == 0$, then

$$\int (d + e x)^m P_q[x] (a + b x + c x^2)^p dx \rightarrow \int (d + e x)^{m+1} \text{PolynomialQuotient}[P_q[x], d + e x, x] (a + b x + c x^2)^p dx$$

Program code:

```
Int[(d_+e_.**x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(d+e*x)^(m+1)*PolynomialQuotient[Pq,d+e*x,x]*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,d+e*x,x],0]
```

```
Int[(d_+e_.**x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(d+e*x)^(m+1)*PolynomialQuotient[Pq,d+e*x,x]*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,d+e*x,x],0]
```

2: $\int (d+ex)^m (a+bx+cx^2)^p (f+gx+hx^2) dx$

when $b e h (m+p+2) + 2 c d h (p+1) - c e g (m+2p+3) = 0 \wedge b d h (p+1) + a e h (m+1) - c e f (m+2p+3) = 0 \wedge m+2p+3 \neq 0$

Derivation: Special case of one step of the Ostrogradskiy-Hermite integration method

Rule 1.2.1.9.2: If $b e h (m+p+2) + 2 c d h (p+1) - c e g (m+2p+3) = 0 \wedge$, then

$$b d h (p+1) + a e h (m+1) - c e f (m+2p+3) = 0 \wedge m+2p+3 \neq 0$$

$$\int (d+ex)^m (a+bx+cx^2)^p (f+gx+hx^2) dx \rightarrow \frac{h (d+ex)^{m+1} (a+bx+cx^2)^{p+1}}{c e (m+2p+3)}$$

Program code:

```
Int[(d_+e_.*x_)^m_.*P2_*(a_+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{f=Coeff[P2,x,0],g=Coeff[P2,x,1],h=Coeff[P2,x,2]},
    h*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(c*e*(m+2*p+3)) /;
    EqQ[b*e*h*(m+p+2)+2*c*d*h*(p+1)-c*e*g*(m+2*p+3),0] && EqQ[b*d*h*(p+1)+a*e*h*(m+1)-c*e*f*(m+2*p+3),0] /;
    FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[P2,x,2] && NeQ[m+2*p+3,0]
```

```
Int[(d_+e_.*x_)^m_.*P2_*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  With[{f=Coeff[P2,x,0],g=Coeff[P2,x,1],h=Coeff[P2,x,2]},
    h*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/(c*e*(m+2*p+3)) /;
    EqQ[2*d*h*(p+1)-e*g*(m+2*p+3),0] && EqQ[a*h*(m+1)-c*f*(m+2*p+3),0] /;
    FreeQ[{a,c,d,e,m,p},x] && PolyQ[P2,x,2] && NeQ[m+2*p+3,0]
```

3: $\int (d+ex)^m P_q(x) (a+bx+cx^2)^p dx$ when $p+2 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.9.3: If $p+2 \in \mathbb{Z}^+$, then

$$\int (d+ex)^m P_q(x) (a+bx+cx^2)^p dx \rightarrow \int \text{ExpandIntegrand}[(d+ex)^m P_q(x) (a+bx+cx^2)^p, x] dx$$

Program code:

```
Int[(d_+e_.**x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*Pq*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

```
Int[(d_+e_.**x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*Pq*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,m},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

4: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac == 0$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4ac == 0$, then $\partial_x \frac{(a+bx+cx^2)^p}{(b+2cx)^{2p}} == 0$

Rule 1.2.1.9.4: If $b^2 - 4ac == 0$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \frac{(a+bx+cx^2)^{\text{FracPart}[p]}}{(4c)^{\text{IntPart}[p]} (b+2cx)^{2\text{FracPart}[p]}} \int (d+ex)^m P_q[x] (b+2cx)^{2p} dx$$

Program code:

```
Int[(d_+e_*x_)^m_*Pq_*(a_+b_*x_+c_*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x)^(2*FracPart[p]))*Int[(d+e*x)^m*Pq*(b+2*c*x)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[b^2-4*a*c,0]
```

$$5. \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0$$

$$1: \int (ex)^m P_q[x] (bx+cx^2)^p dx \text{ when } \text{PolynomialRemainder}[P_q[x], b+cx, x] = 0$$

Derivation: Algebraic simplification

$$\text{Basis: } P_q[x] = \frac{1}{e} \frac{e P_q[x]}{b+cx} (bx+cx^2)$$

Rule 1.2.1.9.5.1: If $\text{PolynomialRemainder}[P_q[x], b+cx, x] = 0$, then

$$\int (ex)^m P_q[x] (bx+cx^2)^p dx \rightarrow e \int (ex)^{m-1} \text{PolynomialQuotient}[P_q[x], b+cx, x] (bx+cx^2)^{p+1} dx$$

Program code:

```
Int[(e.*x_)^m_.*Pq.*(b.*x_+c.*x_^2)^p_.,x_Symbol] :=
  e*Int[(e*x)^(m-1)*PolynomialQuotient[Pq,b+c*x,x]*(b*x+c*x^2)^(p+1),x] /;
FreeQ[{b,c,e,m,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,b+c*x,x],0]
```

$$\mathbf{2:} \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge \text{PolynomialRemainder}[P_q[x], ae+cdx, x] = 0$$

Derivation: Algebraic simplification

Basis: If $cd^2 - bde + ae^2 = 0$, then $(d+ex)(ae+cdx) = de(a+bx+cx^2)$

Rule 1.2.1.9.5.2: If

$b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge \text{PolynomialRemainder}[P_q[x], ae+cdx, x] = 0$,
let $Q_{q-1}[x] \rightarrow \text{PolynomialQuotient}[P_q[x], ae+cdx, x]$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow de \int (d+ex)^{m-1} Q_{q-1}[x] (a+bx+cx^2)^{p+1} dx$$

Program code:

```
Int[(d+_e*_x_)^m_.*Pq_*(a+_b*_x_+c*_x_^2)^p_.,x_Symbol] :=
  d*e*Int[(d+e*x)^(m-1)*PolynomialQuotient[Pq,a*e+c*d*x,x]*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && EqQ[PolynomialRemainder[Pq,a*e+c*d*x,
```

```
Int[(d+_e*_x_)^m_.*Pq_*(a+_c*_x_^2)^p_.,x_Symbol] :=
  d*e*Int[(d+e*x)^(m-1)*PolynomialQuotient[Pq,a*e+c*d*x,x]*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && EqQ[PolynomialRemainder[Pq,a*e+c*d*x,x],0]
```

$$\text{3: } \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge p + \frac{1}{2} \in \mathbb{Z}^- \wedge m > 0$$

Derivation: Algebraic expansion and special quadratic recurrence 2b

Basis: If $cd^2 - bde + ae^2 = 0$, then $(d+ex)(ae+cdx) = de(a+bx+cx^2)$

Rule 1.2.1.9.5.3: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge p + \frac{1}{2} \in \mathbb{Z}^- \wedge m > 0$,

let $Q_{q-1}[x] \rightarrow \text{PolynomialQuotient}[P_q[x], ae+cdx, x]$ and $f \rightarrow \text{PolynomialRemainder}[P_q[x], ae+cdx, x]$, then

$$\begin{aligned} & \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \\ & f \int (d+ex)^m (a+bx+cx^2)^p dx + de \int (d+ex)^{m-1} Q_{q-1}[x] (a+bx+cx^2)^{p+1} dx \rightarrow \\ & \frac{f(2cd-be)(d+ex)^m (a+bx+cx^2)^{p+1}}{e(p+1)(b^2-4ac)} + \\ & \frac{1}{(p+1)(b^2-4ac)} \int (d+ex)^{m-1} (a+bx+cx^2)^{p+1} (de(p+1)(b^2-4ac)Q_{q-1}[x] - f(2cd-be)(m+2p+2)) dx \end{aligned}$$

Program code:

```
Int[(d_+e_.x_)^m_.Pq_*(a_+b_.x_+c_.x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a*e+c*d*x,x], f=PolynomialRemainder[Pq,a*e+c*d*x,x]},
    f*(2*c*d-b*e)*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(e*(p+1)*(b^2-4*a*c)) +
    1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)*
      ExpandToSum[d*e*(p+1)*(b^2-4*a*c)*Q-f*(2*c*d-b*e)*(m+2*p+2),x],x] /;
    FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[p+1/2,0] && GtQ[m,0]
```

```

Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a*e+c*d*x,x], f=PolynomialRemainder[Pq,a*e+c*d*x,x]},
    -d*f*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*a*e*(p+1)) +
    d/(2*a*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1)*ExpandToSum[2*a*e*(p+1)*Q+f*(m+2*p+2),x],x] /;
  FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && ILtQ[p+1/2,0] && GtQ[m,0]

```

4: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge m+q+2p+1 = 0 \wedge m \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Rule 1.2.1.9.5.4: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge m+q+2p+1 = 0 \wedge m \in \mathbb{Z}^-$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \int (a+bx+cx^2)^p \text{ExpandIntegrand}[(d+ex)^m P_q[x], x] dx$$

Program code:

```

Int[(d_+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x+c*x^2)^p,(d+e*x)^m*Pq,x],x] /;
  FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && EqQ[m+Expon[Pq,x]+2*p+1,0] && ILtQ[m,0]

```

```

Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+c*x^2)^p,(d+e*x)^m*Pq,x],x] /;
  FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && EqQ[m+Expon[Pq,x]+2*p+1,0] && ILtQ[m,0]

```

5: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge m+q+2p+1 \neq 0$

Derivation: Algebraic expansion and quadratic recurrence 3a with $A = d$, $B = e$ and $m = m - 1$

Rule 1.2.1.9.5.5: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge m+q+2p+1 \neq 0$, let $f \rightarrow P_q[x, q]$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow$$

$$\int (d+ex)^m \left(P_q[x] - \frac{f}{e^q} (d+ex)^q \right) (a+bx+cx^2)^p dx + \frac{f}{e^q} \int (d+ex)^{m+q} (a+bx+cx^2)^p dx \rightarrow$$

$$\frac{f (d+ex)^{m+q-1} (a+bx+cx^2)^{p+1}}{c e^{q-1} (m+q+2p+1)} + \frac{1}{c e^q (m+q+2p+1)} \int (d+ex)^m (a+bx+cx^2)^p \cdot$$

$$(c e^q (m+q+2p+1) P_q[x] - c f (m+q+2p+1) (d+ex)^q + e f (m+p+q) (d+ex)^{q-2} (bd-2ae+(2cd-be)x)) dx$$

Program code:

```
Int[(d_+e_.**x_)^m_.**Pq_*(a_+b_.**x_+c_.**x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
    f*(d+e*x)^(m+q-1)*(a+b*x+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
    1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+b*x+c*x^2)^p*
      ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q+e*f*(m+p+q)*(d+e*x)^(q-2)*(b*d-2*a*e+(2*c*d-b*e)*x),x],x] /;
  NeQ[m+q+2*p+1,0] /;
  FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[(d_+e_.**x_)^m_.**Pq_*(a_+c_.**x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
    f*(d+e*x)^(m+q-1)*(a+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
    1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+c*x^2)^p*
      ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q-2*e*f*(m+p+q)*(d+e*x)^(q-2)*(a*e-c*d*x),x],x] /;
  NeQ[m+q+2*p+1,0] /;
  FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && Not[IGtQ[m,0]]
```

6: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $cd^2 - bde + ae^2 = 0$, then $a + bx + cx^2 = (d + ex) \left(\frac{a}{d} + \frac{cx}{e} \right)$

Rule 1.2.1.9.5.6: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge p \in \mathbb{Z}$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \int (d+ex)^{m+p} \left(\frac{a}{d} + \frac{cx}{e} \right)^p P_q[x] dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,b,c,d,e,m},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,c,d,e,m},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

7: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $cd^2 - bde + ae^2 = 0$, then $\partial_x \frac{(a+bx+cx^2)^p}{(d+ex)^p \left(\frac{a}{d} + \frac{cx}{e} \right)^p} = 0$

Rule 1.2.1.9.5.7: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 = 0 \wedge p \notin \mathbb{Z}$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \frac{(a+bx+cx^2)^{\text{FracPart}[p]}}{(d+ex)^{\text{FracPart}[p]} \left(\frac{a}{d} + \frac{cx}{e}\right)^{\text{FracPart}[p]}} \int (d+ex)^{m+p} \left(\frac{a}{d} + \frac{cx}{e}\right)^p P_q[x] dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((d+e*x)^FracPart[p]*(a/d+(c*x)/e)^FracPart[p])*Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] && Not[IGtQ[m,0]]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (a+c*x^2)^FracPart[p]/((d+e*x)^FracPart[p]*(a/d+(c*x)/e)^FracPart[p])*Int[(d+e*x)^(m+p)*(a/d+c/e*x)^p*Pq,x] /;
FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && Not[IGtQ[m,0]]
```

6. $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1$

1: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1 \wedge m > 0$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.2.1.9.6.1: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1 \wedge m > 0$,

let $Q_{q-2}[x] \rightarrow \text{PolynomialQuotient}[P_q[x], a+bx+cx^2, x]$ and

$f+gx \rightarrow \text{PolynomialRemainder}[P_q[x], a+bx+cx^2, x]$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow$$

$$\int (d+ex)^m (f+gx) (a+bx+cx^2)^p dx + \int (d+ex)^{m-1} (d+ex) Q_{q-2}[x] (a+bx+cx^2)^{p+1} dx \rightarrow$$

$$\frac{(d+ex)^m (a+bx+cx^2)^{p+1} (fb - 2ag + (2cf - bg)x)}{(p+1)(b^2 - 4ac)} +$$

$$\frac{1}{(p+1)(b^2 - 4ac)} \int (d+ex)^{m-1} (a+bx+cx^2)^{p+1} dx.$$

$$\left((p+1) (b^2 - 4ac) (d+ex) Q_{q-2}[x] + g (2aem + bd(2p+3)) - f (bem + 2cd(2p+3)) - e (2cf - bg) (m+2p+3)x \right) dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x+c*x^2,x],
    f=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,1]},
    (d+e*x)^m*(a+b*x+c*x^2)^(p+1)*(f*b-2*a*g+(2*c*f-b*g)*x)/((p+1)*(b^2-4*a*c)) +
    1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)*
      ExpandToSum[(p+1)*(b^2-4*a*c)*(d+e*x)*Q+g*(2*a*e*m+b*d*(2*p+3))-f*(b*e*m+2*c*d*(2*p+3))-e*(2*c*f-b*g)*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && GtQ[m,0] &&
(IntegerQ[p] || Not[IntegerQ[m]] || Not[RationalQ[a,b,c,d,e]]) &&
Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+c*x^2,x],
    f=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,1]},
    (d+e*x)^m*(a+c*x^2)^(p+1)*(a*g-c*f*x)/(2*a*c*(p+1)) +
    1/(2*a*c*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1)*
      ExpandToSum[2*a*c*(p+1)*(d+e*x)*Q-a*e*g*m+c*d*f*(2*p+3)+c*e*f*(m+2*p+3)*x,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && GtQ[m,0] &&
Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

$$2. \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1 \wedge m \neq 0$$

$$1: \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1 \wedge m \in \mathbb{Z}^-$$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.1.9.6.2.1: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1 \wedge m \in \mathbb{Z}^-$,

let $Q_{m+q-2}[x] \rightarrow \text{PolynomialQuotient}[(d+ex)^m P_q[x], a+bx+cx^2, x]$ and
 $f + gx \rightarrow \text{PolynomialRemainder}[(d+ex)^m P_q[x], a+bx+cx^2, x]$, then

$$\begin{aligned} & \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \\ & \int (f+gx) (a+bx+cx^2)^p dx + \int Q_{m+q-2}[x] (a+bx+cx^2)^{p+1} dx \rightarrow \\ & \frac{(bf - 2ag + (2cf - bg)x)(a+bx+cx^2)^{p+1}}{(p+1)(b^2 - 4ac)} + \\ & \frac{1}{(p+1)(b^2 - 4ac)} \int (d+ex)^m (a+bx+cx^2)^{p+1} ((p+1)(b^2 - 4ac)(d+ex)^{-m} Q_{m+q-2}[x] - (2p+3)(2cf - bg)(d+ex)^{-m}) dx \end{aligned}$$

Program code:

```
Int[(d_+e_*x_)^m_*Pq_*(a_+b_*x_+c_*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[(d+e*x)^m*Pq,a+b*x+c*x^2,x],
    f=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+b*x+c*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+b*x+c*x^2,x],x,1]},
    (b*f-2*a*g+(2*c*f-b*g)*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) +
    1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p+1)*
      ExpandToSum[(p+1)*(b^2-4*a*c)*(d+e*x)^(-m)*Q-(2*p+3)*(2*c*f-b*g)*(d+e*x)^(-m),x],x] /;
  FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && ILtQ[m,0]
```

```

Int[(d+_e_.*x_)^m_.*Pq_*(a+_c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[(d+e*x)^m*Pq,a+c*x^2,x],
    f=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+c*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[(d+e*x)^m*Pq,a+c*x^2,x],x,1]},
    (a*g-c*f*x)*(a+c*x^2)^(p+1)/(2*a*c*(p+1)) +
    1/(2*a*c*(p+1))*Int[(d+e*x)^m*(a+c*x^2)^(p+1)*
      ExpandToSum[2*a*c*(p+1)*(d+e*x)^(-m)*Q+c*f*(2*p+3)*(d+e*x)^(-m),x],x] /;
FreeQ[{a,c,d,e},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && ILtQ[m,0]

```

$$\mathbf{2:} \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1 \wedge m \neq 0$$

Derivation: Algebraic expansion and quadratic recurrence 2b

Rule 1.2.1.9.6.2.2: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge p < -1 \wedge m \neq 0$,

let $Q_{q-2}[x] \rightarrow \text{PolynomialQuotient}[P_q[x], a+bx+cx^2, x]$ and

$f + gx \rightarrow \text{PolynomialRemainder}[P_q[x], a+bx+cx^2, x]$, then

$$\begin{aligned} & \int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \\ & \int (d+ex)^m (f+gx) (a+bx+cx^2)^p dx + \int (d+ex)^m Q_{q-2}[x] (a+bx+cx^2)^{p+1} dx \rightarrow \\ & \left(\left((d+ex)^{m+1} (a+bx+cx^2)^{p+1} (f(bcd - b^2e + 2ace) - ag(2cd - be) + c(f(2cd - be) - g(bd - 2ae))x) \right) / \right. \\ & \quad \left. ((p+1)(b^2 - 4ac)(cd^2 - bde + ae^2)) \right) + \\ & \quad \frac{1}{(p+1)(b^2 - 4ac)(cd^2 - bde + ae^2)} \int (d+ex)^m (a+bx+cx^2)^{p+1} \cdot \\ & \quad ((p+1)(b^2 - 4ac)(cd^2 - bde + ae^2) Q_{q-2}[x] + \\ & \quad f(bcde(2p-m+2) + b^2e^2(p+m+2) - 2c^2d^2(2p+3) - 2ace^2(m+2p+3)) - \\ & \quad g(ae(be - 2cdm + bem) - bd(3cd - be + 2cdp - bep)) + \\ & \quad ce(g(bd - 2ae) - f(2cd - be))(m+2p+4)x) dx \end{aligned}$$

Program code:

```

Int[(d_+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x+c*x^2,x],
    f=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,1]},
    (d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)*(f*(b*c*d-b^2*e+2*a*c*e)-a*g*(2*c*d-b*e)+c*(f*(2*c*d-b*e)-g*(b*d-2*a*e))*x)/
    ((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)) +
    1/((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p+1)*
    ExpandToSum[(p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)*Q+
    f*(b*c*d*e*(2*p-m+2)+b^2*e^2*(p+m+2)-2*c^2*d^2*(2*p+3)-2*a*c*e^2*(m+2*p+3))-
    g*(a*e*(b*e-2*c*d*m+b*e*m)-b*d*(3*c*d-b*e+2*c*d*p-b*e*p))+
    c*e*(g*(b*d-2*a*e)-f*(2*c*d-b*e))*(m+2*p+4)*x,x]] /;
FreeQ[{a,b,c,d,e,m},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] &&
Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]

```

```

Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+c*x^2,x],
    f=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[Pq,a+c*x^2,x],x,1]},
    -(d+e*x)^(m+1)*(a+c*x^2)^(p+1)*(a*(e*f-d*g)+(c*d*f+a*e*g)*x)/(2*a*(p+1)*(c*d^2+a*e^2)) +
    1/(2*a*(p+1)*(c*d^2+a*e^2))*Int[(d+e*x)^m*(a+c*x^2)^(p+1)*
    ExpandToSum[2*a*(p+1)*(c*d^2+a*e^2)*Q+c*d^2*f*(2*p+3)-a*e*(d*g*m-e*f*(m+2*p+3))+e*(c*d*f+a*e*g)*(m+2*p+4)*x,x]] /;
FreeQ[{a,c,d,e,m},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] &&
Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]

```

7: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge m < -1$

Derivation: Algebraic expansion and quadratic recurrence 3b

Rule 1.2.1.9.7: If $b^2 - 4ac \neq 0 \wedge cd^2 - bde + ae^2 \neq 0 \wedge m < -1$,

let $Q_{q-1}[x] \rightarrow \text{PolynomialQuotient}[P_q[x], d+ex, x]$ and $R \rightarrow \text{PolynomialRemainder}[P_q[x], d+ex, x]$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow$$

$$\int (d+ex)^{m+1} Q_{q-1}[x] (a+bx+cx^2)^p dx + R \int (d+ex)^m (a+bx+cx^2)^p dx \rightarrow$$

$$\frac{eR (d+ex)^{m+1} (a+bx+cx^2)^{p+1}}{(m+1) (cd^2 - bde + ae^2)} +$$

$$\frac{1}{(m+1) (cd^2 - bde + ae^2)} \int (d+ex)^{m+1} (a+bx+cx^2)^p dx.$$

$$((m+1) (cd^2 - bde + ae^2) Q_{q-1}[x] + cdR(m+1) - beR(m+p+2) - ceR(m+2p+3)x) dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,d+e*x,x], R=PolynomialRemainder[Pq,d+e*x,x]},
    (e*R*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1))/((m+1)*(c*d^2-b*d*e+a*e^2)) +
    1/((m+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p*
      ExpandToSum[(m+1)*(c*d^2-b*d*e+a*e^2)*Q+c*d*R*(m+1)-b*e*R*(m+p+2)-c*e*R*(m+2*p+3)*x,x],x]] /;
  FreeQ[{a,b,c,d,e,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[m,-1]
```

```
Int[(d_+e_.*x_)^m_*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,d+e*x,x], R=PolynomialRemainder[Pq,d+e*x,x]},
    (e*R*(d+e*x)^(m+1)*(a+c*x^2)^(p+1))/((m+1)*(c*d^2+a*e^2)) +
    1/((m+1)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+1)*(a+c*x^2)^p*
      ExpandToSum[(m+1)*(c*d^2+a*e^2)*Q+c*d*R*(m+1)-c*e*R*(m+2*p+3)*x,x],x]] /;
  FreeQ[{a,c,d,e,p},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && LtQ[m,-1]
```

8: $\int x^m P_q[x] (a+bx^2)^p dx$ when $\neg P_q[x^2] \wedge m+2 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $P_q[x] = \sum_{k=0}^{q/2} P_q[x, 2k] x^{2k} + x \sum_{k=0}^{(q-1)/2} P_q[x, 2k+1] x^{2k}$

Note: This rule transforms $x^m P_q[x]$ into a sum of the form $x^m Q_r[x^2] + x^{m+1} R_s[x^2]$.

Rule 1.2.1.9.8: If $\neg P_q[x^2] \wedge m+2 \in \mathbb{Z}^+$, then

$$\int x^m P_q[x] (a+bx^2)^p dx \rightarrow \int x^m \left(\sum_{k=0}^{\frac{q}{2}} P_q[x, 2k] x^{2k} \right) (a+bx^2)^p dx + \int x^{m+1} \left(\sum_{k=0}^{\frac{q-1}{2}} P_q[x, 2k+1] x^{2k} \right) (a+bx^2)^p dx$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
Module[{q=Expon[Pq,x],k},
Int[x^m*Sum[Coeff[Pq,x,2*k]*x^(2*k),{k,0,q/2}]*(a+b*x^2)^p,x] +
Int[x^(m+1)*Sum[Coeff[Pq,x,2*k+1]*x^(2*k),{k,0,(q-1)/2}]*(a+b*x^2)^p,x] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && Not[PolyQ[Pq,x^2]] && IGtQ[m,-2] && Not[IntegerQ[2*p]]
```

9: $\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx$ when $b^2-4ac \neq 0 \wedge cd^2-bde+ae^2 \neq 0 \wedge m+q+2p+1 \neq 0$

Derivation: Algebraic expansion and quadratic recurrence 3a with $A = d$, $B = e$ and $m = m - 1$

Rule 1.2.1.9.9: If $b^2-4ac \neq 0 \wedge cd^2-bde+ae^2 \neq 0 \wedge m+q+2p+1 \neq 0$, let $f \rightarrow P_q[x, q]$, then

$$\int (d+ex)^m P_q[x] (a+bx+cx^2)^p dx \rightarrow \int (d+ex)^m \left(P_q[x] - \frac{f}{e^q} (d+ex)^q \right) (a+bx+cx^2)^p dx + \frac{f}{e^q} \int (d+ex)^{m+q} (a+bx+cx^2)^p dx \rightarrow$$

$$\frac{f(d+ex)^{m+q-1} (a+bx+cx^2)^{p+1}}{ce^{q-1} (m+q+2p+1)} + \frac{1}{ce^q (m+q+2p+1)} \int (d+ex)^m (a+bx+cx^2)^p (ce^q (m+q+2p+1) P_q[x] - cf(m+q+2p+1) (d+ex)^q - f(d+ex)^{q-2} (bde(p+1) + ae^2(m+q-1) - cd^2(m+q+2p+1) - e(2cd-be)(m+q+p)x)) dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
    f*(d+e*x)^(m+q-1)*(a+b*x+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
    1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+b*x+c*x^2)^p*ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q-
      f*(d+e*x)^(q-2)*(b*d*e*(p+1)+a*e^2*(m+q-1)-c*d^2*(m+q+2*p+1)-e*(2*c*d-b*e)*(m+q+p)*x),x],x] /;
  GtQ[q,1] && NeQ[m+q+2*p+1,0] /;
  FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],f=Coeff[Pq,x,Expon[Pq,x]]},
    f*(d+e*x)^(m+q-1)*(a+c*x^2)^(p+1)/(c*e^(q-1)*(m+q+2*p+1)) +
    1/(c*e^q*(m+q+2*p+1))*Int[(d+e*x)^m*(a+c*x^2)^p*ExpandToSum[c*e^q*(m+q+2*p+1)*Pq-c*f*(m+q+2*p+1)*(d+e*x)^q-
      f*(d+e*x)^(q-2)*(a*e^2*(m+q-1)-c*d^2*(m+q+2*p+1)-2*c*d*e*(m+q+p)*x),x],x] /;
  GtQ[q,1] && NeQ[m+q+2*p+1,0] /;
  FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] && Not[EqQ[d,0] && True] &&
  Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

10: $\int (d+e x)^m P_q[x] (a+b x+c x^2)^p dx$ when $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0$

Derivation: Algebraic expansion

$$\text{Basis: } (d+e x)^m P_q[x] == \frac{P_q[x,q] (d+e x)^{m+q}}{e^q} + \frac{(d+e x)^m (e^q P_q[x] - P_q[x,q] (d+e x)^q)}{e^q}$$

Rule 1.2.1.9.10: If $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0$, then

$$\int (d+e x)^m P_q[x] (a+b x+c x^2)^p dx \rightarrow \frac{P_q[x,q]}{e^q} \int (d+e x)^{m+q} (a+b x+c x^2)^p dx + \frac{1}{e^q} \int (d+e x)^m (a+b x+c x^2)^p (e^q P_q[x] - P_q[x,q] (d+e x)^q) dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    Coeff[Pq,x,q]/e^q*Int[(d+e*x)^(m+q)*(a+b*x+c*x^2)^p,x] +
    1/e^q*Int[(d+e*x)^m*(a+b*x+c*x^2)^p*ExpandToSum[e^q*Pq-Coeff[Pq,x,q]*(d+e*x)^q,x],x] /;
  FreeQ[{a,b,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  Not[IGtQ[m,0] && RationalQ[a,b,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```

```
Int[(d_+e_.*x_)^m_.*Pq_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    Coeff[Pq,x,q]/e^q*Int[(d+e*x)^(m+q)*(a+c*x^2)^p,x] +
    1/e^q*Int[(d+e*x)^m*(a+c*x^2)^p*ExpandToSum[e^q*Pq-Coeff[Pq,x,q]*(d+e*x)^q,x],x] /;
  FreeQ[{a,c,d,e,m,p},x] && PolyQ[Pq,x] && NeQ[c*d^2+a*e^2,0] &&
  Not[IGtQ[m,0] && RationalQ[a,c,d,e] && (IntegerQ[p] || ILtQ[p+1/2,0])]
```