Rules for integrands of the form $(f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n$

1. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d$

   1. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n>0$

      1. $\int x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n>0$

         1: $\int \frac{x\,(a+b\,\text{ArcSinh}[c\,x])^n}{d+e\,x^2}\,dx$ when $e == c^2\,d \land n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $e == c^2\,d$, then $\frac{x}{d+e\,x^2} == \frac{1}{e}\,\text{Subst}[\text{Tanh}[x],\,x,\,\text{ArcSinh}[c\,x]]\,\partial_x\text{ArcSinh}[c\,x]$

Basis: If $c^2\,d+e == 0$, then $\frac{x}{d+e\,x^2} == \frac{1}{e}\,\text{Subst}[\text{Coth}[x],\,x,\,\text{ArcCosh}[c\,x]]\,\partial_x\text{ArcCosh}[c\,x]$

Note: If $n\in\mathbb{Z}^+$, then $(a+b\,x)^n\,\text{Tanh}[x]$ is integrable in closed-form.

Rule: If $e == c^2\,d \land n\in\mathbb{Z}^+$, then

$$\int \frac{x\,(a+b\,\text{ArcSinh}[c\,x])^n}{d+e\,x^2}\,dx \rightarrow \frac{1}{e}\,\text{Subst}\left[\int (a+b\,x)^n\,\text{Tanh}[x]\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

Program code:

```
Int[x_*(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  1/e*Subst[Int[(a+b*x)^n*Tanh[x],x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[n,0]
```

```
Int[x_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  1/e*Subst[Int[(a+b*x)^n*Coth[x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

2. $\int x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n>0 \land p \neq -1$

1: $\int x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n>0 \land p \neq -1 \land (p \in \mathbb{Z} \lor d>0)$

Derivation: Integration by parts

Rule: If $e == c^2\,d \land n>0 \land p \neq -1 \land (p \in \mathbb{Z} \lor d>0)$, then

$$\int x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$
$$\frac{(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{2\,e\,(p+1)}-\frac{b\,n\,d^p}{2\,c\,(p+1)}\int(1+c^2\,x^2)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
(* Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*e*(p+1)) -
  b*n*d^p/(2*c*(p+1))*Int[(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && NeQ[p,-1] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e*(p+1)) -
  b*n*(-d)^p/(2*c*(p+1))*Int[(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && NeQ[p,-1] && IntegerQ[p]
```

```
(* Int[x_*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e1*e2*(p+1)) -
  b*n*(-d1*d2)^(p-1/2)/(2*c*(p+1))*Int[(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && IntegerQ[p+1/2] && (GtQ[d1,0] && LtQ[d2,0]) *)
```

2: $\int x\left(d+e x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n>0 \land p \neq -1$

Derivation: Integration by parts and piecewise constant extraction

Basis: If $e == c^2 d$, then $\partial_x \frac{(d+e x^2)^p}{(1+c^2 x^2)^p} == 0$

Rule: If $e == c^2 d \land n>0 \land p \neq -1$, then

$$\int x\left(d+e x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{\left(d+e x^2\right)^{p+1}(a+b\,\text{ArcSinh}[c\,x])^n}{2 e\,(p+1)}-\frac{b\,n\,d^{\text{IntPart}[p]}\left(d+e x^2\right)^{\text{FracPart}[p]}}{2 c\,(p+1)\left(1+c^2 x^2\right)^{\text{FracPart}[p]}}\int\left(1+c^2 x^2\right)^{p+\frac{1}{2}}(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*e*(p+1)) -
  b*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(2*c*(p+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && NeQ[p,-1]
```

```
Int[x_*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e1*e2*(p+1)) -
  b*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(2*c*(p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && NeQ[p,-1] && IntegerQ[p+1/2]
```

```
Int[x_*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e1*e2*(p+1)) -
  b*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(2*c*(p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && NeQ[p,-1]
```

2. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\land n>0\land m+2\,p+3==0$

1: $\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{x\,(d+e\,x^2)}\,dx$ when $e == c^2\,d\land n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $e == c^2\,d$, then $\frac{1}{x\,(d+e\,x^2)} == \frac{1}{d}\,\text{Subst}\left[\frac{1}{\text{Cosh}[x]\,\text{Sinh}[x]},\,x,\,\text{ArcSinh}[c\,x]\right]\,\partial_x\text{ArcSinh}[c\,x]$

Rule: If $e == c^2\,d\land n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{x\,(d+e\,x^2)}\,dx\;\longrightarrow\;\frac{1}{d}\,\text{Subst}\left[\int \frac{(a+b\,x)^n}{\text{Cosh}[x]\,\text{Sinh}[x]}\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(x_*(d_+e_.*x_^2)),x_Symbol] :=
  1/d*Subst[Int[(a+b*x)^n/(Cosh[x]*Sinh[x]),x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(x_*(d_+e_.*x_^2)),x_Symbol] :=
  -1/d*Subst[Int[(a+b*x)^n/(Cosh[x]*Sinh[x]),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

2. $\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n>0 \land m+2p+3==0 \land m \neq -1$

   1: $\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n>0 \land m+2p+3==0 \land m \neq -1 \land (p \in \mathbb{Z} \lor d>0)$

Derivation: Integration by parts

Basis: If $m+2p+3==0$, then $(f x)^m (d+e x^2)^p == \partial_x \frac{(f x)^{m+1} (d+e x^2)^{p+1}}{d f (m+1)}$

Rule: If $e == c^2 d \land n>0 \land m+2p+3==0 \land m \neq -1 \land (p \in \mathbb{Z} \lor d>0)$, then

$$\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n \, dx \longrightarrow$$
$$\frac{(f x)^{m+1} (d+e x^2)^{p+1} (a+b \text{ArcSinh}[c x])^n}{d f (m+1)} - \frac{b c n d^p}{f (m+1)} \int (f x)^{m+1} (1+c^2 x^2)^{p+\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n-1} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(d*f*(m+1)) -
  b*c*n*d^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[m,-1] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(d*f*(m+1)) +
  b*c*n*(-d)^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[m,-1] && IntegerQ[p]
```

```
(* Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(d1*d2*f*(m+1)) +
  b*c*n*(-d1*d2)^p/(f*(m+1))*Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && EqQ[m+2*p+3,0] &&
  NeQ[m,-1] && IntegerQ[p+1/2] && (GtQ[d1,0] && LtQ[d2,0]) *)
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e==c^2\,d\,\wedge\,n>0\,\wedge\,m+2\,p+3==0\,\wedge\,m\neq-1$

Derivation: Integration by parts and piecewise constant extraction

Basis: If $m+2\,p+3==0$, then $(f\,x)^m\,(d+e\,x^2)^p==\partial_x\frac{(f\,x)^{m+1}\,(d+e\,x^2)^{p+1}}{d\,f\,(m+1)}$

Basis: If $e==c^2\,d$, then $\partial_x\frac{(d+e\,x^2)^p}{(1+c^2\,x^2)^p}==0$

Rule: If $e==c^2\,d\,\wedge\,n>0\,\wedge\,m+2\,p+3==0\,\wedge\,m\neq-1$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\,\longrightarrow$$
$$\frac{(f\,x)^{m+1}\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{d\,f\,(m+1)}-\frac{b\,c\,n\,d^{\text{IntPart}[p]}\,(d+e\,x^2)^{\text{FracPart}[p]}}{f\,(m+1)\,(1+c^2\,x^2)^{\text{FracPart}[p]}}\int (f\,x)^{m+1}\,(1+c^2\,x^2)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(d*f*(m+1)) -
  b*c*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(f*(m+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[m,-1]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(d1*d2*f*(m+1)) +
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(f*(m+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[m,-1] && IntegerQ[p+1/2]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(d1*d2*f*(m+1)) +
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(f*(m+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[m,-1]
```

3. $\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right) dx$ when $e == c^2 d \land p \in \mathbb{Z}^+$

1. $\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right) dx$ when $e == c^2 d \land p \in \mathbb{Z}^+ \land \frac{m-1}{2} \in \mathbb{Z}^-$

1: $\int \frac{\left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)}{x}\,dx$ when $e == c^2 d \land p \in \mathbb{Z}^+$

Derivation: Inverted integration by parts

Rule: If $e == c^2 d \land p \in \mathbb{Z}^+$, then

$$\int \frac{\left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)}{x}\,dx \longrightarrow$$
$$\frac{\left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)}{2\,p} - \frac{b\,c\,d^p}{2\,p} \int \left(1+c^2\,x^2\right)^{p-\frac{1}{2}} dx + d \int \frac{\left(d+e\,x^2\right)^{p-1} \left(a+b\,\text{ArcSinh}[c\,x]\right)}{x}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])/x_,x_Symbol] :=
  (d+e*x^2)^p*(a+b*ArcSinh[c*x])/(2*p) -
  b*c*d^p/(2*p)*Int[(1+c^2*x^2)^(p-1/2),x] +
  d*Int[(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])/x,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])/x_,x_Symbol] :=
  (d+e*x^2)^p*(a+b*ArcCosh[c*x])/(2*p) -
  b*c*(-d)^p/(2*p)*Int[(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2),x] +
  d*Int[(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])/x,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e = c^2\,d\,\wedge\,p\in\mathbb{Z}^+\wedge\,\frac{m+1}{2}\in\mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $e = c^2\,d\,\wedge\,p\in\mathbb{Z}^+\wedge\,\frac{m+1}{2}\in\mathbb{Z}^-$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx \;\rightarrow$$
$$\frac{(f\,x)^{m+1}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])}{f\,(m+1)}-$$
$$\frac{b\,c\,d^p}{f\,(m+1)}\int (f\,x)^{m+1}\,(1+c^2\,x^2)^{p-\frac{1}{2}}\,dx-\frac{2\,e\,p}{f^2\,(m+1)}\int (f\,x)^{m+2}\,(d+e\,x^2)^{p-1}\,(a+b\,\text{ArcSinh}[c\,x])\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])/(f*(m+1)) -
  b*c*d^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2),x] -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && IGtQ[p,0] && ILtQ[(m+1)/2,0]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])/(f*(m+1)) -
  b*c*(-d)^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2),x] -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && IGtQ[p,0] && ILtQ[(m+1)/2,0]
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e == c^2\,d \land p \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $e == c^2\,d \land p \in \mathbb{Z}^+$, let $u = \int (f\,x)^m\,(d+e\,x^2)^p\,dx$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcSinh}[c\,x]) - b\,c\int \frac{u}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

4. $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x]) \, dx$ when $e == c^2 d \land p-\frac{1}{2} \in \mathbb{Z} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2 p+3}{2} \in \mathbb{Z}^-\right)$

1: $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x]) \, dx$ when $e == c^2 d \land p-\frac{1}{2} \in \mathbb{Z} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2 p+3}{2} \in \mathbb{Z}^-\right) \land p \neq -\frac{1}{2} \land d > 0$

Derivation: Integration by parts

Note: If $p-\frac{1}{2} \in \mathbb{Z} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2 p+3}{2} \in \mathbb{Z}^-\right)$, then $\int x^m \left(1+c^2 x^2\right)^p dx$ is an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.

Rule: If $e == c^2 d \land p-\frac{1}{2} \in \mathbb{Z} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2 p+3}{2} \in \mathbb{Z}^-\right) \land p \neq -\frac{1}{2} \land d > 0$, let $u = \int x^m \left(1+c^2 x^2\right)^p dx$, then

$$\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x]) \, dx \to d^p u (a+b \operatorname{ArcSinh}[c x]) - b c d^p \int \frac{u}{\sqrt{1+c^2 x^2}} \, dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(1+c^2*x^2)^p,x]},
  Dist[d^p*(a+b*ArcSinh[c*x]),u,x] - b*c*d^p*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IntegerQ[p-1/2] && (IGtQ[(m+1)/2,0] || ILtQ[(m+2*p+3)/2,0]) && NeQ[p,-1/2] && GtQ[d,0]
```

```
Int[x_^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(1+c*x)^p*(-1+c*x)^p,x]},
  Dist[(-d1*d2)^p*(a+b*ArcCosh[c*x]),u,x] - b*c*(-d1*d2)^p*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2] &&
  (IGtQ[(m+1)/2,0] || ILtQ[(m+2*p+3)/2,0]) && NeQ[p,-1/2] && GtQ[d1,0] && LtQ[d2,0]
```

2: $\int x^m \left(d+e\,x^2\right)^p (a+b\,\mathrm{ArcSinh}[c\,x])\,dx$ when $e == c^2 d \land p+\frac{1}{2} \in \mathbb{Z}^+ \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2p+3}{2} \in \mathbb{Z}^-\right)$

Derivation: Integration by parts and piecewise constant extraction

Note: If $p+\frac{1}{2} \in \mathbb{Z} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2p+3}{2} \in \mathbb{Z}^-\right)$, then $\int x^m \left(1+c^2 x^2\right)^p dx$ is an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.

Rule: If $e == c^2 d \land p+\frac{1}{2} \in \mathbb{Z}^+ \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2p+3}{2} \in \mathbb{Z}^-\right)$, let $u = \int x^m \left(1+c^2 x^2\right)^p dx$, then

$$\int x^m \left(d+e\,x^2\right)^p (a+b\,\mathrm{ArcSinh}[c\,x])\,dx \to (a+b\,\mathrm{ArcSinh}[c\,x]) \int x^m \left(d+e\,x^2\right)^p dx - \frac{b\,c\,d^{p-\frac{1}{2}}\sqrt{d+e\,x^2}}{\sqrt{1+c^2 x^2}} \int \frac{u}{\sqrt{1+c^2 x^2}}\,dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(1+c^2*x^2)^p,x]},
  (a+b*ArcSinh[c*x])*Int[x^m*(d+e*x^2)^p,x] -
  b*c*d^(p-1/2)*Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p+1/2,0] && (IGtQ[(m+1)/2,0] || ILtQ[(m+2*p+3)/2,0])
```

```
Int[x_^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(1+c*x)^p*(-1+c*x)^p,x]},
  (a+b*ArcCosh[c*x])*Int[x^m*(d1+e1*x)^p*(d2+e2*x)^p,x] -
  b*c*(-d1*d2)^(p-1/2)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(Sqrt[1+c*x]*Sqrt[-1+c*x])*
    Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[p+1/2,0] && (IGtQ[(m+1)/2,0] || ILtQ[(m+2*p+3)/2,0])
```

5. $\int (f x)^m (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p > 0$

   1. $\int (f x)^m (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p > 0 \land m < -1$

      1: $\int (f x)^m (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p > 0 \land m < -1 \land (p \in \mathbb{Z} \lor d > 0)$

Derivation: Inverted integration by parts

Rule: If $e == c^2 d \land n > 0 \land p > 0 \land m < -1 \land (p \in \mathbb{Z} \lor d > 0)$, then

$$\int (f x)^m (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} (d+e x^2)^p (a+b \operatorname{ArcSinh}[c x])^n}{f (m+1)} -$$

$$\frac{2 e p}{f^2 (m+1)} \int (f x)^{m+2} (d+e x^2)^{p-1} (a+b \operatorname{ArcSinh}[c x])^n \, dx - \frac{b c n d^p}{f (m+1)} \int (f x)^{m+1} (1+c^2 x^2)^{p-\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(f*(m+1)) -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n*d^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d)^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1] && IntegerQ[p]
```

```
(* Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  2*e1*e2*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^p/(f*(m+1))*Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1] &&
  IntegerQ[p-1/2] && (GtQ[d1,0] && LtQ[d2,0]) *)
```

2. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n>0 \land p>0 \land m<-1$

1: $\int (f\,x)^m\,\sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n>0 \land m<-1$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $e == c^2 d \land n>0 \land m<-1$, then

$$\int (f\,x)^m\,\sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f\,x)^{m+1}\,\sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n}{f\,(m+1)}-$$

$$\frac{b\,c\,n\,\sqrt{d+e\,x^2}}{f\,(m+1)\,\sqrt{1+c^2\,x^2}}\int (f\,x)^{m+1}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx-\frac{c^2\,\sqrt{d+e\,x^2}}{f^2\,(m+1)\,\sqrt{1+c^2\,x^2}}\int \frac{(f\,x)^{m+2}\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/(f*(m+1)) -
  b*c*n*Sqrt[d+e*x^2]/(f*(m+1)*Sqrt[1+c^2*x^2])*Int[(f*x)^(m+1)*(a+b*ArcSinh[c*x])^(n-1),x] -
  c^2*Sqrt[d+e*x^2]/(f^2*(m+1)*Sqrt[1+c^2*x^2])*Int[(f*x)^(m+2)*(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[m,-1]
```

```
Int[(f_.*x_)^m_*Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  b*c*n*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(f*(m+1)*Sqrt[1+c*x]*Sqrt[-1+c*x])*
    Int[(f*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1),x] -
  c^2*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(f^2*(m+1)*Sqrt[1+c*x]*Sqrt[-1+c*x])*
    Int[((f*x)^(m+2)*(a+b*ArcCosh[c*x])^n)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && LtQ[m,-1]
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e=c^2\,d\land n>0\land p>0\land m<-1$

Derivation: Inverted integration by parts

Rule: If $e=c^2\,d\land n>0\land p>0\land m<-1$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\,\longrightarrow$$
$$\frac{(f\,x)^{m+1}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n}{f\,(m+1)}-\frac{2\,e\,p}{f^2\,(m+1)}\int (f\,x)^{m+2}\,(d+e\,x^2)^{p-1}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx-$$
$$\frac{b\,c\,n\,d^{\text{IntPart}[p]}\,(d+e\,x^2)^{\text{FracPart}[p]}}{f\,(m+1)\,(1+c^2\,x^2)^{\text{FracPart}[p]}}\int (f\,x)^{m+1}\,(1+c^2\,x^2)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(f*(m+1)) -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(f*(m+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  2*e1*e2*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^(p-1/2)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(f*(m+1)*Sqrt[1+c*x]*Sqrt[-1+c*x])*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1] && IntegerQ[p-1/2]
```

2. $\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\mathtt{ArcSinh}[c\,x]\right)^n\,dx$ when $e == c^2 d \land n>0 \land p>0 \land m \not< -1$

1: $\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\mathtt{ArcSinh}[c\,x]\right)^n\,dx$ when $e == c^2 d \land n>0 \land p>0 \land m \not< -1 \land (p \in \mathbb{Z} \lor d>0)$

Derivation: Inverted integration by parts

Rule: If $e == c^2 d \land n>0 \land p>0 \land m \not< -1 \land (p \in \mathbb{Z} \lor d>0)$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\mathtt{ArcSinh}[c\,x]\right)^n\,dx \longrightarrow$$

$$\frac{(f\,x)^{m+1} \left(d+e\,x^2\right)^p \left(a+b\,\mathtt{ArcSinh}[c\,x]\right)^n}{f\,(m+2\,p+1)}+$$

$$\frac{2\,d\,p}{m+2\,p+1}\int (f\,x)^m \left(d+e\,x^2\right)^{p-1} \left(a+b\,\mathtt{ArcSinh}[c\,x]\right)^n\,dx-\frac{b\,c\,n\,d^p}{f\,(m+2\,p+1)}\int (f\,x)^{m+1} \left(1+c^2\,x^2\right)^{p-\frac{1}{2}} \left(a+b\,\mathtt{ArcSinh}[c\,x]\right)^{n-1}\,dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(f*(m+2*p+1)) +
  2*d*p/(m+2*p+1)*Int[(f*x)^m*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n*d^p/(f*(m+2*p+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0] && Not[LtQ[m,-1]] && (IntegerQ[p] || GtQ[d,0]) && (RationalQ[m] ||
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n/(f*(m+2*p+1)) +
  2*d*p/(m+2*p+1)*Int[(f*x)^m*(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d)^p/(f*(m+2*p+1))*Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0] && Not[LtQ[m,-1]] && IntegerQ[p] && (RationalQ[m] || EqQ[n,1])
```

2. $\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)^n \,dx$ when $e = c^2 d \land n > 0 \land p > 0 \land m \nless -1$

1: $\int (f\,x)^m \sqrt{d+e\,x^2}\,\left(a+b\,\text{ArcSinh}[c\,x]\right)^n \,dx$ when $e = c^2 d \land n > 0 \land m \nless -1$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $e = c^2 d \land n > 0 \land m \nless -1$, then

$$\int (f\,x)^m \sqrt{d+e\,x^2}\,\left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx \to \frac{(f\,x)^{m+1}\sqrt{d+e\,x^2}\,\left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{f\,(m+2)} -$$
$$\frac{b\,c\,n\,\sqrt{d+e\,x^2}}{f\,(m+2)\,\sqrt{1+c^2 x^2}} \int (f\,x)^{m+1}\left(a+b\,\text{ArcSinh}[c\,x]\right)^{n-1} dx + \frac{\sqrt{d+e\,x^2}}{(m+2)\,\sqrt{1+c^2 x^2}} \int \frac{(f\,x)^m \left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{\sqrt{1+c^2 x^2}} dx$$

Program code:

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/(f*(m+2)) -
  b*c*n*Sqrt[d+e*x^2]/(f*(m+2)*Sqrt[1+c^2*x^2])*Int[(f*x)^(m+1)*(a+b*ArcSinh[c*x])^(n-1),x] +
  Sqrt[d+e*x^2]/((m+2)*Sqrt[1+c^2*x^2])*Int[(f*x)^m*(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && Not[LtQ[m,-1]] && (RationalQ[m] || EqQ[n,1])
```

```
Int[(f_.*x_)^m_*Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/(f*(m+2)) -
  b*c*n*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(f*(m+2)*Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[(f*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1),x] -
  Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/((m+2)*Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[(f*x)^m*(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && Not[LtQ[m,-1]] && (RationalQ[m] || EqQ[n,1])
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)^n\,dx$ when $e == c^2 d \land n > 0 \land p > 0 \land m \nless -1$

Derivation: Inverted integration by parts

Rule: If $e == c^2 d \land n > 0 \land p > 0 \land m \nless -1$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)^n\,dx \longrightarrow$$

$$\frac{(f\,x)^{m+1} \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)^n}{f\,(m+2\,p+1)} + \frac{2\,d\,p}{m+2\,p+1} \int (f\,x)^m \left(d+e\,x^2\right)^{p-1} \left(a+b\,\text{ArcSinh}[c\,x]\right)^n\,dx -$$
$$\frac{b\,c\,n\,d^{\text{IntPart}[p]} \left(d+e\,x^2\right)^{\text{FracPart}[p]}}{f\,(m+2\,p+1) \left(1+c^2\,x^2\right)^{\text{FracPart}[p]}} \int (f\,x)^{m+1} \left(1+c^2\,x^2\right)^{p-\frac{1}{2}} \left(a+b\,\text{ArcSinh}[c\,x]\right)^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(f*(m+2*p+1)) +
  2*d*p/(m+2*p+1)*Int[(f*x)^m*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(f*(m+2*p+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0] && Not[LtQ[m,-1]] && (RationalQ[m] || EqQ[n,1])
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(f*(m+2*p+1)) +
  2*d1*d2*p/(m+2*p+1)*Int[(f*x)^m*(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^(p-1/2)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(f*(m+2*p+1)*Sqrt[1+c*x]*Sqrt[-1+c*x])*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && GtQ[p,0] && Not[LtQ[m,-1]] &&
  IntegerQ[p-1/2] && (RationalQ[m] || EqQ[n,1])
```

6. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,m<-1\,\wedge\,m\in\mathbb{Z}$

1: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,m<-1\,\wedge\,m\in\mathbb{Z}\,\wedge\,(p\in\mathbb{Z}\,\vee\,d>0)$

Rule: If $e == c^2\,d\,\wedge\,n>0\,\wedge\,m<-1\,\wedge\,m\in\mathbb{Z}\,\wedge\,(p\in\mathbb{Z}\,\vee\,d>0)$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\,\longrightarrow$$
$$\frac{(f\,x)^{m+1}\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{d\,f\,(m+1)}-$$
$$\frac{c^2\,(m+2\,p+3)}{f^2\,(m+1)}\int (f\,x)^{m+2}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx-\frac{b\,c\,n\,d^p}{f\,(m+1)}\int (f\,x)^{m+1}\,(1+c^2\,x^2)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Programcode:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(d*f*(m+1)) -
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n*d^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[m,-1] && IntegerQ[m] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(d*f*(m+1)) +
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] +
  b*c*n*(-d)^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[m,-1] && IntegerQ[m] && IntegerQ[p]
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e = c^2\,d \land n>0 \land m<-1 \land m \in \mathbb{Z}$

Rule: If $e = c^2\,d \land n>0 \land m<-1 \land m \in \mathbb{Z}$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f\,x)^{m+1} \left(d+e\,x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c\,x])^n}{d\,f\,(m+1)} - \frac{c^2\,(m+2\,p+3)}{f^2\,(m+1)} \int (f\,x)^{m+2} \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx -$$
$$\frac{b\,c\,n\,d^{\text{IntPart}[p]} \left(d+e\,x^2\right)^{\text{FracPart}[p]}}{f\,(m+1)\,\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}} \int (f\,x)^{m+1} \left(1+c^2\,x^2\right)^{p+\frac{1}{2}} (a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Programcode:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(d*f*(m+1)) -
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(f*(m+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[m,-1] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(d1*d2*f*(m+1)) +
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] +
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(f*(m+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && LtQ[m,-1] && IntegerQ[m] && IntegerQ[p+1/2]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(d1*d2*f*(m+1)) +
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] +
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(f*(m+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && LtQ[m,-1] && IntegerQ[m]
```

7. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,p<-1$

1. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,p<-1\,\wedge\,m>1$

1: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,p<-1\,\wedge\,m>1\,\wedge\,(p\in\mathbb{Z}\,\vee\,d>0)$

Derivation: Integration by parts

Basis: $x\,(d+e\,x^2)^p == \partial_x \frac{(d+e\,x^2)^{p+1}}{2\,e\,(p+1)}$

Rule: If $e == c^2\,d\,\wedge\,n>0\,\wedge\,p<-1\,\wedge\,m>1\,\wedge\,(p\in\mathbb{Z}\,\vee\,d>0)$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \to$$

$$\frac{f\,(f\,x)^{m-1}\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{2\,e\,(p+1)} -$$

$$\frac{f^2\,(m-1)}{2\,e\,(p+1)}\int (f\,x)^{m-2}\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx - \frac{b\,f\,n\,d^p}{2\,c\,(p+1)}\int (f\,x)^{m-1}\,(1+c^2\,x^2)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*e*(p+1)) -
  f^2*(m-1)/(2*e*(p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] -
  b*f*n*d^p/(2*c*(p+1))*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && GtQ[m,1] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e*(p+1)) -
  f^2*(m-1)/(2*e*(p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n*(-d)^p/(2*c*(p+1))*Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && GtQ[m,1] && IntegerQ[p]
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n \,dx$ when $e == c^2 d \land n>0 \land p<-1 \land m>1$

Derivation: Integration by parts

Basis: $x\left(d+e\,x^2\right)^p == \partial_x \frac{\left(d+e\,x^2\right)^{p+1}}{2\,e\,(p+1)}$

Rule: If $e == c^2 d \land n>0 \land p<-1 \land m>1$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n \,dx \longrightarrow$$

$$\frac{f\,(f\,x)^{m-1}\left(d+e\,x^2\right)^{p+1}(a+b\,\text{ArcSinh}[c\,x])^n}{2\,e\,(p+1)}-\frac{f^2\,(m-1)}{2\,e\,(p+1)}\int (f\,x)^{m-2}\left(d+e\,x^2\right)^{p+1}(a+b\,\text{ArcSinh}[c\,x])^n\,dx-$$
$$\frac{b\,f\,n\,d^{\text{IntPart}[p]}\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{2\,c\,(p+1)\left(1+c^2 x^2\right)^{\text{FracPart}[p]}}\int (f\,x)^{m-1}\left(1+c^2 x^2\right)^{p+\frac{1}{2}}(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*e*(p+1)) -
  f^2*(m-1)/(2*e*(p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] -
  b*f*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(2*c*(p+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m-1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && GtQ[m,1]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e1*e2*(p+1)) -
  f^2*(m-1)/(2*e1*e2*(p+1))*Int[(f*x)^(m-2)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(2*c*(p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m-1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && LtQ[p,-1] && GtQ[m,1] && IntegerQ[p+1/2]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e1*e2*(p+1)) -
  f^2*(m-1)/(2*e1*e2*(p+1))*Int[(f*x)^(m-2)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(2*c*(p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && LtQ[p,-1] && Not[IntegerQ[p]] && GtQ[m,1]
```

2. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e==c^2\,d\land n>0\land p<-1\land m\ngtr 1$

   1: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e==c^2\,d\land n>0\land p<-1\land m\ngtr 1\land(p\in\mathbb{Z}\lor d>0)$

Rule: If $e==c^2\,d\land n>0\land p<-1\land m\ngtr 1\land(p\in\mathbb{Z}\lor d>0)$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\,\longrightarrow$$
$$-\frac{(f\,x)^{m+1}\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{2\,d\,f\,(p+1)}+$$
$$\frac{m+2\,p+3}{2\,d\,(p+1)}\int (f\,x)^m\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx+\frac{b\,c\,n\,d^p}{2\,f\,(p+1)}\int (f\,x)^{m+1}\,(1+c^2\,x^2)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*d*f*(p+1)) +
  (m+2*p+3)/(2*d*(p+1))*Int[(f*x)^m*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] +
  b*c*n*d^p/(2*f*(p+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] && (IntegerQ[p] || GtQ[d,0]) &&
  (IntegerQ[m] || IntegerQ[p] || EqQ[n,1]) *)
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d*f*(p+1)) +
  (m+2*p+3)/(2*d*(p+1))*Int[(f*x)^m*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d)^p/(2*f*(p+1))*Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] && IntegerQ[p]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b\,\text{ArcSinh}[c x])^n\,dx$ when $e == c^2 d \land n>0 \land p<-1 \land m \not> 1$

Rule: If $e == c^2 d \land n>0 \land p<-1 \land m \not> 1$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b\,\text{ArcSinh}[c x])^n\,dx \longrightarrow$$
$$-\frac{(f x)^{m+1} \left(d+e x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c x])^n}{2 d f (p+1)}+\frac{m+2 p+3}{2 d (p+1)} \int (f x)^m \left(d+e x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c x])^n\,dx+$$
$$\frac{b c n d^{\text{IntPart}[p]} \left(d+e x^2\right)^{\text{FracPart}[p]}}{2 f (p+1) \left(1+c^2 x^2\right)^{\text{FracPart}[p]}} \int (f x)^{m+1} \left(1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b\,\text{ArcSinh}[c x])^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*d*f*(p+1)) +
  (m+2*p+3)/(2*d*(p+1))*Int[(f*x)^m*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] +
  b*c*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(2*f*(p+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] && (IntegerQ[m] || IntegerQ[p] || EqQ[n,1])
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d1*d2*f*(p+1)) +
  (m+2*p+3)/(2*d1*d2*(p+1))*Int[(f*x)^m*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(2*f*(p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] &&
  (IntegerQ[m] || EqQ[n,1]) && IntegerQ[p+1/2]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d1*d2*f*(p+1)) +
  (m+2*p+3)/(2*d1*d2*(p+1))*Int[(f*x)^m*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(2*f*(p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] &&
  (IntegerQ[m] || IntegerQ[p] || EqQ[n,1])
```

8. $\int \frac{(f x)^m (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $e = c^2 d \land n > 0$

   1. $\int \frac{(f x)^m (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $e = c^2 d \land n > 0 \land m > 1$

      1: $\int \frac{(f x)^m (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $e = c^2 d \land n > 0 \land m > 1 \land d > 0$

Rule: If $e = c^2 d \land n > 0 \land m > 1 \land d > 0$, then

$$\int \frac{(f x)^m (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx \longrightarrow$$

$$\frac{f (f x)^{m-1} \sqrt{d+e x^2} (a+b \operatorname{ArcSinh}[c x])^n}{e m} - \frac{b f n}{c m \sqrt{d}} \int (f x)^{m-1} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx - \frac{f^2 (m-1)}{c^2 m} \int \frac{(f x)^{m-2} (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/(e*m) -
  b*f*n*Sqrt[1+c^2*x^2]/(c*m*Sqrt[d+e*x^2])*Int[(f*x)^(m-1)*(a+b*ArcSinh[c*x])^(n-1),x] -
  f^2*(m-1)/(c^2*m)*Int[((f*x)^(m-2)*(a+b*ArcSinh[c*x])^n)/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[m,1] && GtQ[d,0] && IntegerQ[m] *)
```

```
(* Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/(e1*e2*m) +
  b*f*n*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(c*d1*d2*m*Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[(f*x)^(m-1)*(a+b*ArcCosh[c*x])^(n-1),x] +
  f^2*(m-1)/(c^2*m)*Int[(f*x)^(m-2)*(a+b*ArcCosh[c*x])^n/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && GtQ[m,1] && IntegerQ[m] *)
```

2: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \land n>0 \land m>1$

Rule: If $e == c^2\,d \land n>0 \land m>1$, then

$$\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow$$
$$\frac{f\,(f\,x)^{m-1}\,\sqrt{d+e\,x^2}\,(a+b\,\text{ArcSinh}[c\,x])^n}{e\,m} -$$
$$\frac{b\,f\,n\,\sqrt{1+c^2\,x^2}}{c\,m\,\sqrt{d+e\,x^2}}\int (f\,x)^{m-1}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx - \frac{f^2\,(m-1)}{c^2\,m}\int \frac{(f\,x)^{m-2}\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/(e*m) -
  b*f*n*Sqrt[1+c^2*x^2]/(c*m*Sqrt[d+e*x^2])*Int[(f*x)^(m-1)*(a+b*ArcSinh[c*x])^(n-1),x] -
  f^2*(m-1)/(c^2*m)*Int[((f*x)^(m-2)*(a+b*ArcSinh[c*x])^n)/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[m,1] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/(e1*e2*m) +
  b*f*n*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(c*d1*d2*m*Sqrt[1+c*x]*Sqrt[-1+c*x])*Int[(f*x)^(m-1)*(a+b*ArcCosh[c*x])^(n-1),x] +
  f^2*(m-1)/(c^2*m)*Int[(f*x)^(m-2)*(a+b*ArcCosh[c*x])^n/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && GtQ[m,1] && IntegerQ[m]
```

2: $\int \frac{x^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \land d>0 \land n\in\mathbb{Z}^+ \land m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $e == c^2\,d \land d>0 \land m\in\mathbb{Z}$, then $\frac{x^m}{\sqrt{d+e\,x^2}} == \frac{1}{c^{m+1}\sqrt{d}}\,\text{Subst}[\text{Sinh}[x]^m,\,x,\,\text{ArcSinh}[c\,x]]\,\partial_x \text{ArcSinh}[c\,x]$

Note: If $n\in\mathbb{Z}^+$, then $(a+b\,x)^n\,\text{Sinh}[x]$ is integrable in closed-form.

Rule: If $e == c^2\,d \land d>0 \land n\in\mathbb{Z}^+ \land m\in\mathbb{Z}$, then

$$\int \frac{x^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \to \frac{1}{c^{m+1}\sqrt{d}}\,\text{Subst}\left[\int (a+b\,x)^n\,\text{Sinh}[x]^m\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

Program code:

```
Int[x_^m_*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(c^(m+1)*Sqrt[d])*Subst[Int[(a+b*x)^n*Sinh[x]^m,x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[d,0] && IGtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  1/(c^(m+1)*Sqrt[-d1*d2])*Subst[Int[(a+b*x)^n*Cosh[x]^m,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[n,0] && GtQ[d1,0] && LtQ[d2,0] && IntegerQ[m]
```

3: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \land d>0 \land m\notin\mathbb{Z}$

Rule: If $e == c^2\,d \land d>0 \land m\notin\mathbb{Z}$, then

$$\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])}{\sqrt{d+e\,x^2}}\,dx \to$$

$$\frac{1}{\sqrt{d}\; f\,(m+1)} (f\,x)^{m+1} \left(a + b\,\text{ArcSinh}[c\,x]\right) \text{Hypergeometric2F1}\left[\frac{1}{2}, \frac{1+m}{2}, \frac{3+m}{2}, -c^2\,x^2\right] - \left(b\,c\,(f\,x)^{m+2}\,\text{HypergeometricPFQ}\left[\left\{1, 1+\frac{m}{2}, 1+\frac{m}{2}\right\}, \left\{\frac{3}{2}+\frac{m}{2}, 2+\frac{m}{2}\right\}, -c^2\,x^2\right]\right) \Big/ \left(\sqrt{d}\; f^2\,(m+1)\,(m+2)\right)$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^(m+1)*(a+b*ArcSinh[c*x])*Hypergeometric2F1[1/2,(1+m)/2,(3+m)/2,-c^2*x^2]/(Sqrt[d]*f*(m+1)) -
  b*c*(f*x)^(m+2)*HypergeometricPFQ[{1,1+m/2,1+m/2},{3/2+m/2,2+m/2},-c^2*x^2]/(Sqrt[d]*f^2*(m+1)*(m+2)) /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[d,0] && Not[IntegerQ[m]]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  (f*x)^(m+1)*Sqrt[1-c^2*x^2]*(a+b*ArcCosh[c*x])*Hypergeometric2F1[1/2,(1+m)/2,(3+m)/2,c^2*x^2]/
    (f*(m+1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]) +
  b*c*(f*x)^(m+2)*HypergeometricPFQ[{1,1+m/2,1+m/2},{3/2+m/2,2+m/2},c^2*x^2]/(Sqrt[-d1*d2]*f^2*(m+1)*(m+2)) /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[d1,0] && LtQ[d2,0] && Not[IntegerQ[m]]
```

4: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,d\not>0$

Derivation: Piecewise constant extraction

Basis: If $e == c^2\,d$, then $\partial_x \frac{\sqrt{1+c^2\,x^2}}{\sqrt{d+e\,x^2}} == 0$

Rule: If $e == c^2\,d\,\wedge\,n>0\,\wedge\,d\not>0$, then

$$\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{\sqrt{1+c^2\,x^2}}{\sqrt{d+e\,x^2}}\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(f*x)^m*(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && Not[GtQ[d,0]] && (IntegerQ[m] || EqQ[n,1])
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  Sqrt[1+c*x]*Sqrt[-1+c*x]/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x])*Int[(f*x)^m*(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && Not[GtQ[d1,0] && LtQ[d2,0]] && (IntegerQ[m] || E
```

9. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,m>1\,\wedge\,m+2\,p+1\neq 0$

1: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,m>1\,\wedge\,m+2\,p+1\neq 0\,\wedge\,(p\in\mathbb{Z}\,\vee\,d>0)$

Rule: If $e == c^2\,d\,\wedge\,n>0\,\wedge\,m>1\,\wedge\,m+2\,p+1\neq 0\,\wedge\,(p\in\mathbb{Z}\,\vee\,d>0)$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$
$$\frac{f\,(f\,x)^{m-1}\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{e\,(m+2\,p+1)} -$$

$$\frac{f^2\,(m-1)}{c^2\,(m+2\,p+1)}\int(f\,x)^{m-2}\,\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx-\frac{b\,f\,n\,d^p}{c\,(m+2\,p+1)}\int(f\,x)^{m-1}\,\left(1+c^2\,x^2\right)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(e*(m+2*p+1)) -
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] -
  b*f*n*d^p/(c*(m+2*p+1))*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[m,1] && NeQ[m+2*p+1,0] && (IntegerQ[p] || GtQ[d,0]) && IntegerQ[m] *)
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(e*(m+2*p+1)) +
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n*(-d)^p/(c*(m+2*p+1))*Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[m,1] && NeQ[m+2*p+1,0] && IntegerQ[p] && IntegerQ[m]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n>0 \land m>1 \land m+2 p+1 \neq 0$

Rule: If $e == c^2 d \land n>0 \land m>1 \land m+2 p+1 \neq 0$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{f (f x)^{m-1} \left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcSinh}[c x])^n}{e (m+2 p+1)}-\frac{f^2 (m-1)}{c^2 (m+2 p+1)} \int (f x)^{m-2} \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx-$$
$$\frac{b f n\, d^{\operatorname{IntPart}[p]} \left(d+e x^2\right)^{\operatorname{FracPart}[p]}}{c (m+2 p+1) \left(1+c^2 x^2\right)^{\operatorname{FracPart}[p]}} \int (f x)^{m-1} \left(1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(e*(m+2*p+1)) -
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] -
  b*f*n*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(c*(m+2*p+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m-1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[m,1] && NeQ[m+2*p+1,0] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(e1*e2*(m+2*p+1)) +
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(c*(m+2*p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m-1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && GtQ[m,1] && NeQ[m+2*p+1,0] && IntegerQ[m] && Int
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(e1*e2*(m+2*p+1)) +
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(c*(m+2*p+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[n,0] && GtQ[m,1] && NeQ[m+2*p+1,0] && IntegerQ[m]
```

2. $\int (f x)^m (d + e x^2)^p (a + b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1$

1. $\int (f x)^m (d + e x^2)^p (a + b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1 \land m + 2 p + 1 == 0$

1: $\int (f x)^m (d + e x^2)^p (a + b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n < -1 \land m + 2 p + 1 == 0 \land (p \in \mathbb{Z} \lor d > 0)$

Derivation: Integration by parts

Basis: $\frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n+1)}$

Rule: If $e == c^2 d \land n < -1 \land m + 2 p + 1 == 0 \land (p \in \mathbb{Z} \lor d > 0)$, then

$$\int (f x)^m (d + e x^2)^p (a + b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{d^p (f x)^m (1 + c^2 x^2)^{p+\frac{1}{2}} (a + b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n + 1)} - \frac{f m d^p}{b c (n + 1)} \int (f x)^{m-1} (1 + c^2 x^2)^{p-\frac{1}{2}} (a + b \operatorname{ArcSinh}[c x])^{n+1} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  d^p*(f*x)^m*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m*d^p/(b*c*(n+1))*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[e,c^2*d] && LtQ[n,-1] && EqQ[m+2*p+1,0] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m*(-d)^p/(b*c*(n+1))*Int[(f*x)^(m-1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && EqQ[m+2*p+1,0] && IntegerQ[p]
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n < -1 \land m+2p+1 == 0$

Derivation: Integration by parts

Basis: $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Rule: If $e == c^2 d \land n < -1 \land m+2p+1 == 0$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \to$$
$$\frac{(f\,x)^m \sqrt{1+c^2 x^2}\left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)} -$$
$$\frac{f\,m\,d^{\text{IntPart}[p]}\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{b\,c\,(n+1)\left(1+c^2 x^2\right)^{\text{FracPart}[p]}} \int (f\,x)^{m-1}\left(1+c^2 x^2\right)^{p-\frac{1}{2}} (a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c^2*x^2]*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(b*c*(n+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m-1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[e,c^2*d] && LtQ[n,-1] && EqQ[m+2*p+1,0]
```

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(b*c*(n+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m-1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && LtQ[n,-1] && EqQ[m+2*p+1,0] && IntegerQ[p-1/2]
```

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(b*c*(n+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m-1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && LtQ[n,-1] && EqQ[m+2*p+1,0]
```

2: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d\,\wedge\,n<-1\,\wedge\,d>0$

Derivation: Integration by parts

Basis: If $e == c^2\,d\,\wedge\,d>0$, then $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)}$

Rule: If $e == c^2\,d\,\wedge\,n<-1\,\wedge\,d>0$, then

$$\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \to \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,\sqrt{d}\,(n+1)} - \frac{f\,m}{b\,c\,\sqrt{d}\,(n+1)}\int (f\,x)^{m-1}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^m*(a+b*ArcSinh[c*x])^(n+1)/(b*c*Sqrt[d]*(n+1)) -
  f*m/(b*c*Sqrt[d]*(n+1))*Int[(f*x)^(m-1)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && LtQ[n,-1] && GtQ[d,0]
```

```
Int[(f_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  (f*x)^m*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  (f*m)/(b*c*Sqrt[-d1*d2]*(n+1))*Int[(f*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && LtQ[n,-1] && GtQ[d1,0] && LtQ[d2,0]
```

x: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \land n < -1 \land d \not> 0$

Derivation: Piecewise constant extraction

Basis: If $e == c^2\,d$, then $\partial_x \frac{\sqrt{1+c^2\,x^2}}{\sqrt{d+e\,x^2}} == 0$

Rule: If $e == c^2\,d \land n < -1 \land d \not> 0$, then

$$\int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \to \frac{\sqrt{1+c^2\,x^2}}{\sqrt{d+e\,x^2}} \int \frac{(f\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(f*x)^m*(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && LtQ[n,-1] && Not[GtQ[d,0]] *)
```

```
(* Int[(f_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  Sqrt[1+c*x]*Sqrt[-1+c*x]/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x])*Int[(f*x)^m*(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && LtQ[n,-1] && Not[GtQ[d1,0] && LtQ[d2,0]] *)
```

3. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n < -1 \land m+3 \in \mathbb{Z}^+ \land 2\,p \in \mathbb{Z}^+$

1: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n < -1 \land m+3 \in \mathbb{Z}^+ \land 2\,p \in \mathbb{Z}^+ \land (p \in \mathbb{Z} \lor d > 0)$

Derivation: Integration by parts

Basis: $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Rule: If $e == c^2\,d \land n < -1 \land m+3 \in \mathbb{Z}^+ \land 2\,p \in \mathbb{Z}^+ \land (p \in \mathbb{Z} \lor d > 0)$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{d^p (f x)^m \left(1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n+1)} -$$

$$\frac{f m d^p}{b c (n+1)} \int (f x)^{m-1} \left(1+c^2 x^2\right)^{p-\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n+1} \, dx - \frac{c d^p (m+2 p+1)}{b f (n+1)} \int (f x)^{m+1} \left(1+c^2 x^2\right)^{p-\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n+1} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  d^p*(f*x)^m*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m*d^p/(b*c*(n+1))*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] -
  c*d^p*(m+2*p+1)/(b*f*(n+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && LtQ[n,-1] && IGtQ[m,-3] && IGtQ[2*p,0] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m*(-d)^p/(b*c*(n+1))*Int[(f*x)^(m-1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] -
  c*(-d)^p*(m+2*p+1)/(b*f*(n+1))*Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && IGtQ[m,-3] && IGtQ[p,0]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \wedge n < -1 \wedge m+3 \in \mathbb{Z}^+ \wedge 2 p \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n+1)}$

Rule: If $e == c^2 d \wedge n < -1 \wedge m+3 \in \mathbb{Z}^+ \wedge 2 p \in \mathbb{Z}^+$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^m \sqrt{1+c^2 x^2} \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n+1)} -$$

$$\frac{f\,m\,d^{\text{IntPart}[p]}\,\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{b\,c\,(n+1)\,\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}}\int (f\,x)^{m-1}\,\left(1+c^2\,x^2\right)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx\,-$$
$$\frac{c\,(m+2\,p+1)\,d^{\text{IntPart}[p]}\,\left(d+e\,x^2\right)^{\text{FracPart}[p]}}{b\,f\,(n+1)\,\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}}\int (f\,x)^{m+1}\,\left(1+c^2\,x^2\right)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c^2*x^2]*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(b*c*(n+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m-1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] -
  c*(m+2*p+1)*d^IntPart[p]*(d+e*x^2)^FracPart[p]/(b*f*(n+1)*(1+c^2*x^2)^FracPart[p])*
    Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && LtQ[n,-1] && IGtQ[m,-3] && IGtQ[2*p,0]
```

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(b*c*(n+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m-1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] -
  c*(m+2*p+1)*(-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/(b*f*(n+1)*(1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && LtQ[n,-1] && IGtQ[m,-3] && IGtQ[p+1/2,0]
```

3. $\int x^m \left(d+e x^2\right)^p \left(a+b\,\mathrm{ArcSinh}[c\,x]\right)^n dx$ when $e == c^2 d \land 2p \in \mathbb{Z} \land p > -1 \land m \in \mathbb{Z}^+$

1: $\int x^m \left(d+e x^2\right)^p \left(a+b\,\mathrm{ArcSinh}[c\,x]\right)^n dx$ when $e == c^2 d \land 2p \in \mathbb{Z} \land p > -1 \land m \in \mathbb{Z}^+ \land (p \in \mathbb{Z} \lor d > 0)$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{c}\,\mathrm{Subst}\left[F\left[\frac{\mathrm{Sinh}[x]}{c}\right]\mathrm{Cosh}[x],\ x,\ \mathrm{ArcSinh}[c\,x]\right]\partial_x \mathrm{ArcSinh}[c\,x]$

Basis: If $e == c^2 d \land m \in \mathbb{Z} \land (p \in \mathbb{Z} \lor d > 0)$, then
$x^m \left(d+e x^2\right)^p == \frac{d^p}{c^{m+1}}\,\mathrm{Subst}\left[\mathrm{Sinh}[x]^m\,\mathrm{Cosh}[x]^{2p+1},\ x,\ \mathrm{ArcSinh}[c\,x]\right]\partial_x \mathrm{ArcSinh}[c\,x]$

Rule: If $e == c^2 d \land 2p \in \mathbb{Z} \land p > -1 \land m \in \mathbb{Z}^+ \land (p \in \mathbb{Z} \lor d > 0)$, then

$$\int x^m \left(d+e x^2\right)^p \left(a+b\,\mathrm{ArcSinh}[c\,x]\right)^n dx \longrightarrow \frac{d^p}{c^{m+1}}\,\mathrm{Subst}\left[\int (a+b\,x)^n\,\mathrm{Sinh}[x]^m\,\mathrm{Cosh}[x]^{2p+1}\,dx,\ x,\ \mathrm{ArcSinh}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  d^p/c^(m+1)*Subst[Int[(a+b*x)^n*Sinh[x]^m*Cosh[x]^(2*p+1),x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && IntegerQ[2*p] && GtQ[p,-1] && IGtQ[m,0] && (IntegerQ[p] || GtQ[d,0])
```

```
Int[x_^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^p/c^(m+1)*Subst[Int[(a+b*x)^n*Cosh[x]^m*Sinh[x]^(2*p+1),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && IGtQ[p,0] && IGtQ[m,0]
```

```
Int[x_^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^p/c^(m+1)*Subst[Int[(a+b*x)^n*Cosh[x]^m*Sinh[x]^(2*p+1),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p+1/2] && GtQ[p,-1] && IGtQ[m,0] && (GtQ[d1,0] && LtQ
```

2: $\int x^m \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx$ when $e == c^2 d \land 2p \in \mathbb{Z} \land p > -1 \land m \in \mathbb{Z}^+ \land \neg\left(p \in \mathbb{Z} \lor d > 0\right)$

Derivation: Piecewise constant extraction

Basis: If $e == c^2 d$, then $\partial_x \frac{\left(d+e\,x^2\right)^p}{\left(1+c^2\,x^2\right)^p} == 0$

Rule: If $e == c^2 d \land 2p \in \mathbb{Z} \land p > -1 \land m \in \mathbb{Z}^+ \land \neg\left(p \in \mathbb{Z} \lor d > 0\right)$, then

$$\int x^m \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx \to \frac{d^{\text{IntPart}[p]} \left(d+e\,x^2\right)^{\text{FracPart}[p]}}{\left(1+c^2\,x^2\right)^{\text{FracPart}[p]}} \int x^m \left(1+c^2\,x^2\right)^p \left(a+b\,\text{ArcSinh}[c\,x]\right)^n dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  d^IntPart[p]*(d+e*x^2)^FracPart[p]/(1+c^2*x^2)^FracPart[p]*Int[x^m*(1+c^2*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && IntegerQ[2*p] && GtQ[p,-1] && IGtQ[m,0] && Not[(IntegerQ[p] || GtQ[d,0])]
```

```
Int[x_^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/((1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[x^m*(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[2*p] && GtQ[p,-1] && IGtQ[m,0] &&
  Not[IntegerQ[p] || GtQ[d1,0] && LtQ[d2,0]]
```

4: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,d>0\,\wedge\,p+\frac{1}{2}\in\mathbb{Z}^+\,\wedge\,\frac{m+1}{2}\notin\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $e == c^2\,d\,\wedge\,d>0\,\wedge\,p+\frac{1}{2}\in\mathbb{Z}^+\,\wedge\,\frac{m+1}{2}\notin\mathbb{Z}^+$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\;\longrightarrow\;\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,\text{ExpandIntegrand}\Big[(f\,x)^m\,(d+e\,x^2)^{p+\frac{1}{2}},\,x\Big]\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n/Sqrt[d+e*x^2],(f*x)^m*(d+e*x^2)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[e,c^2*d] && GtQ[d,0] && IGtQ[p+1/2,0] && Not[IGtQ[(m+1)/2,0]] && (EqQ[m,-1] || EqQ[m,-2])
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),(f*x)^m*(d1+e1*x)^(p+1/2)*(d2+e2*x)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[p+1/2,0] && Not[IGtQ[(m+1)/
  (EqQ[m,-1] || EqQ[m,-2])
```

2. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e \neq c^2\,d$

0: $\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,\text{ArcCosh}[c\,x])\,dx$ when $c^2\,d+e\neq 0\,\land\,m\neq -1\,\land\,m\neq -3$

Derivation: Integration by parts

Note: This rule can be removed when integrands of the form $(d+e\,x)^m\,(f+g\,x)^m\,(a+c\,x^2)^p$ when $e\,f+d\,g==0$ are integrated without first resorting to piecewise constant extraction.

Rule: If $c^2\,d+e\neq 0\,\land\,m\neq -1\,\land\,m\neq -3$, then

$$\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,\text{ArcCosh}[c\,x])\,dx\,\longrightarrow$$
$$\frac{d\,(f\,x)^{m+1}\,(a+b\,\text{ArcCosh}[c\,x])}{f\,(m+1)}+\frac{e\,(f\,x)^{m+3}\,(a+b\,\text{ArcCosh}[c\,x])}{f^3\,(m+3)}-\frac{b\,c}{f\,(m+1)\,(m+3)}\int\frac{(f\,x)^{m+1}\,(d\,(m+3)+e\,(m+1)\,x^2)}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  d*(f*x)^(m+1)*(a+b*ArcCosh[c*x])/(f*(m+1)) +
  e*(f*x)^(m+3)*(a+b*ArcCosh[c*x])/(f^3*(m+3)) -
  b*c/(f*(m+1)*(m+3))*Int[(f*x)^(m+1)*(d*(m+3)+e*(m+1)*x^2)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[c^2*d+e,0] && NeQ[m,-1] && NeQ[m,-3]
```

1: $\int x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e\neq c^2\,d\,\land\,p\neq -1$

Derivation: Integration by parts

Basis:: If $p\neq -1$, then $x\,(d+e\,x^2)^p = \partial_x \frac{(d+e\,x^2)^{p+1}}{2\,e\,(p+1)}$

Rule: If $e\neq c^2\,d\,\land\,p\neq -1$, then

$$\int x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx \to \frac{(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])}{2\,e\,(p+1)} - \frac{b\,c}{2\,e\,(p+1)}\int\frac{(d+e\,x^2)^{p+1}}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])/(2*e*(p+1)) - b*c/(2*e*(p+1))*Int[(d+e*x^2)^(p+1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[e,c^2*d] && NeQ[p,-1]
```

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])/(2*e*(p+1)) - b*c/(2*e*(p+1))*Int[(d+e*x^2)^(p+1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[c^2*d+e,0] && NeQ[p,-1]
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e\neq c^2\,d\,\land\,p\in\mathbb{Z}\,\land\,\left(p>0\,\lor\,\frac{m-1}{2}\in\mathbb{Z}^+\land\,m+p\leq 0\right)$

Derivation: Integration by parts

Note: If $\frac{m-1}{2}\in\mathbb{Z}^+\land\,p\in\mathbb{Z}^-\land\,m+p\geq 0$, then $\int(f\,x)^m\,(d+e\,x^2)^p$ is a rational function.

Rule: If $e\neq c^2\,d\,\land\,p\in\mathbb{Z}\,\land\,\left(p>0\,\lor\,\frac{m-1}{2}\in\mathbb{Z}^+\land\,m+p\leq 0\right)$, let $u=\int(f\,x)^m\,(d+e\,x^2)^p\,dx$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\mathrm{ArcSinh}[c\,x])\,dx \,\longrightarrow\, u\,(a+b\,\mathrm{ArcSinh}[c\,x])-b\,c\int\frac{u}{\sqrt{1+c^2\,x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[e,c^2*d] && IntegerQ[p] && (GtQ[p,0] || IGtQ[(m-1)/2,0] && LeQ[m+p,0])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[c^2*d+e,0] && IntegerQ[p] && (GtQ[p,0] || IGtQ[(m-1)/2,0] && LeQ[m+p,0])
```

3: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\mathrm{ArcSinh}[c\,x])^n\,dx$ when $e\neq c^2\,d\,\wedge\,n\in\mathbb{Z}^+\wedge\,p\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $e\neq c^2\,d\,\wedge\,n\in\mathbb{Z}^+\wedge\,p\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\mathrm{ArcSinh}[c\,x])^n\,dx \,\longrightarrow\, \int (a+b\,\mathrm{ArcSinh}[c\,x])^n\,\mathrm{ExpandIntegrand}\left[(f\,x)^m\,(d+e\,x^2)^p,\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n,(f*x)^m*(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[e,c^2*d] && IGtQ[n,0] && IntegerQ[p] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n,(f*x)^m*(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[c^2*d+e,0] && IGtQ[n,0] && IntegerQ[p] && IntegerQ[m]
```

X: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$

Rule:

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,n,p},x]
```

Rules for integrands of the form $(h\,x)^m\,(d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n$

1: $\int (h\,x)^m\,(d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e\,f+d\,g==0\,\land\,c^2\,d^2+e^2==0\,\land\,(p\mid q)\in\mathbb{Z}+\frac{1}{2}\,\land\,p-q\geq 0\,\land\,d>0\,\land\,\frac{g}{e}<0$

Derivation: Algebraic expansion

Basis: If $e\,f+d\,g==0\,\land\,c^2\,d^2+e^2==0\,\land\,d>0\,\land\,\frac{g}{e}<0$, then
$(d+e\,x)^p\,(f+g\,x)^q==\left(-\frac{d^2\,g}{e}\right)^q\,(d+e\,x)^{p-q}\,\left(1+c^2\,x^2\right)^q$

Rule: If $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land (p \mid q) \in \mathbb{Z} + \frac{1}{2} \land p - q \geq 0 \land d > 0 \land \frac{g}{e} < 0$, then

$$\int (h x)^m (d + e x)^p (f + g x)^q (a + b \operatorname{ArcSinh}[c x])^n \, dx \rightarrow \left(-\frac{d^2 g}{e}\right)^q \int (h x)^m (d + e x)^{p-q} (1 + c^2 x^2)^q (a + b \operatorname{ArcSinh}[c x])^n \, dx$$

Program code:

```
Int[(h_.*x_)^m_.*(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^q*Int[(h*x)^m*(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0] && GtQ[d,0] && LtQ[g/e,0]
```

2: $\int (h x)^m (d + e x)^p (f + g x)^q (a + b \operatorname{ArcSinh}[c x])^n \, dx$ when $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land (p \mid q) \in \mathbb{Z} + \frac{1}{2} \land p - q \geq 0 \land \neg \left(d > 0 \land \frac{g}{e} < 0\right)$

Derivation: Piecewise constant extraction

Basis: If $e f + d g == 0 \land c^2 d^2 + e^2 == 0$, then $\partial_x \frac{(d+e x)^q (f+g x)^q}{(1+c^2 x^2)^q} == 0$

Rule: If $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land (p \mid q) \in \mathbb{Z} + \frac{1}{2} \land p - q \geq 0 \land \neg \left(d > 0 \land \frac{g}{e} < 0\right)$, then

$$\int (h x)^m (d + e x)^p (f + g x)^q (a + b \operatorname{ArcSinh}[c x])^n \, dx \rightarrow$$

$$\frac{\left(-\frac{d^2 g}{e}\right)^{\operatorname{IntPart}[q]} (d + e x)^{\operatorname{FracPart}[q]} (f + g x)^{\operatorname{FracPart}[q]}}{(1 + c^2 x^2)^{\operatorname{FracPart}[q]}} \int (h x)^m (d + e x)^{p-q} (1 + c^2 x^2)^q (a + b \operatorname{ArcSinh}[c x])^n \, dx$$

Program code:

```
Int[(h_.*x_)^m_.*(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^IntPart[q]*(d+e*x)^FracPart[q]*(f+g*x)^FracPart[q]/(1+c^2*x^2)^FracPart[q]*
    Int[(h*x)^m*(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^IntPart[p]*(d+e*x^2)^FracPart[p]/((1+c*x)^FracPart[p]*(-1+c*x)^FracPart[p])*
    Int[(f*x)^m*(1+c*x)^p*(-1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[c^2*d+e,0] && Not[IntegerQ[p]]
```