Rules for integrands of the form $\left(a\,x^j+b\,x^n\right)^p$

1: $\int\left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\land j\neq n\land j\,p-n+j+1==0$

Derivation: Generalized binomial recurrence 2a with $m=0$ and $j\,p-n+j+1==0$

Rule: If $p\notin\mathbb{Z}\land j\neq n\land j\,p-n+j+1==0$, then

$$\int\left(a\,x^j+b\,x^n\right)^p\,dx\;\longrightarrow\;\frac{\left(a\,x^j+b\,x^n\right)^{p+1}}{b\,(n-j)\,(p+1)\,x^{n-1}}$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^n)^(p+1)/(b*(n-j)(p+1)*x^(n-1)) /;
FreeQ[{a,b,j,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && EqQ[j*p-n+j+1,0]
```

2. $\int \left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{n\,p+n-j+1}{n-j}\in\mathbb{Z}^-$

1: $\int \left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{n\,p+n-j+1}{n-j}\in\mathbb{Z}^-\land\,p<-1$

Derivation: Generalized binomial recurrence 2b with $m=0$

Note: This rule increments $\frac{n\,p+n-j+1}{n-j}$ by 1 thus driving it to 0.

Rule: If $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{n\,p+n-j+1}{n-j}\in\mathbb{Z}^-\land\,p<-1\,\land\,(j\in\mathbb{Z}\,\lor\,c>0)$, then

$$\int \left(a\,x^j+b\,x^n\right)^p\,dx\,\longrightarrow\,-\frac{\left(a\,x^j+b\,x^n\right)^{p+1}}{a\,(n-j)\,(p+1)\,x^{j-1}}+\frac{n\,p+n-j+1}{a\,(n-j)\,(p+1)}\int\frac{\left(a\,x^j+b\,x^n\right)^{p+1}}{x^j}\,dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  -(a*x^j+b*x^n)^(p+1)/(a*(n-j)*(p+1)*x^(j-1)) +
  (n*p+n-j+1)/(a*(n-j)*(p+1))*Int[(a*x^j+b*x^n)^(p+1)/x^j,x] /;
FreeQ[{a,b,j,n},x] && Not[IntegerQ[p]] && NeQ[n,j] && ILtQ[Simplify[(n*p+n-j+1)/(n-j)],0] && LtQ[p,-1]
```

2: $\int \left(a\,x^j+b\,x^n\right)^p dx$ when $p\notin\mathbb{Z}\land j\neq n\land \frac{n\,p+n-j+1}{n-j}\in\mathbb{Z}^-\land j\,p+1\neq 0$

Derivation: Generalized binomial recurrence 3b with $m=0$

Note: This rule increments $\frac{n\,p+n-j+1}{n-j}$ by 1 thus driving it to 0.

Rule: If $p\notin\mathbb{Z}\land j\neq n\land \frac{n\,p+n-j+1}{n-j}\in\mathbb{Z}^-\land j\,p+1\neq 0$, then

$$\int \left(a\,x^j+b\,x^n\right)^p dx \longrightarrow \frac{\left(a\,x^j+b\,x^n\right)^{p+1}}{a\,(j\,p+1)\,x^{j-1}}-\frac{b\,(n\,p+n-j+1)}{a\,(j\,p+1)}\int x^{n-j}\left(a\,x^j+b\,x^n\right)^p dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^n)^(p+1)/(a*(j*p+1)*x^(j-1)) -
  b*(n*p+n-j+1)/(a*(j*p+1))*Int[x^(n-j)*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,j,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && ILtQ[Simplify[(n*p+n-j+1)/(n-j)],0] && NeQ[j*p+1,0]
```

4. $\int (a\,x^j + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n$

1. $\int (a\,x^j + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land p > 0$

1: $\int (a\,x^j + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land p > 0 \land j\,p + 1 < 0$

Derivation: Generalized binomial recurrence 1a with $m = 0$

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land p > 0 \land j\,p + 1 < 0$, then

$$\int (a\,x^j + b\,x^n)^p\,dx \longrightarrow \frac{x\,(a\,x^j + b\,x^n)^p}{j\,p + 1} - \frac{b\,(n - j)\,p}{j\,p + 1} \int x^n\,(a\,x^j + b\,x^n)^{p-1}\,dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  x*(a*x^j+b*x^n)^p/(j*p+1) -
  b*(n-j)*p/(j*p+1)*Int[x^n*(a*x^j+b*x^n)^(p-1),x] /;
FreeQ[{a,b},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && GtQ[p,0] && LtQ[j*p+1,0]
```

2: $\int (a\,x^j + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land p > 0 \land n\,p + 1 \neq 0$

Derivation: Generalized binomial recurrence 1b with $m = 0$

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land p > 0 \land n\,p + 1 \neq 0$, then

$$\int (a\,x^j + b\,x^n)^p\,dx \longrightarrow \frac{x\,(a\,x^j + b\,x^n)^p}{n\,p + 1} + \frac{a\,(n - j)\,p}{n\,p + 1} \int x^j\,(a\,x^j + b\,x^n)^{p-1}\,dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  x*(a*x^j+b*x^n)^p/(n*p+1) +
  a*(n-j)*p/(n*p+1)*Int[x^j*(a*x^j+b*x^n)^(p-1),x] /;
FreeQ[{a,b},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && GtQ[p,0] && NeQ[n*p+1,0]
```

2. $\int (a\,x^j + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land p < -1$

1: $\int (a\,x^j + b\,x^n)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land p < -1 \land j\,p + 1 > n - j$

Derivation: Generalized binomial recurrence 2a with $m = 0$

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land p < -1 \land j\,p + 1 > n - j$, then

$$\int (a\,x^j + b\,x^n)^p\,dx \to \frac{(a\,x^j + b\,x^n)^{p+1}}{b\,(n-j)\,(p+1)\,x^{n-1}} - \frac{j\,p - n + j + 1}{b\,(n-j)\,(p+1)} \int \frac{(a\,x^j + b\,x^n)^{p+1}}{x^n}\,dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^n)^(p+1)/(b*(n-j)*(p+1)*x^(n-1)) -
  (j*p-n+j+1)/(b*(n-j)*(p+1))*Int[(a*x^j+b*x^n)^(p+1)/x^n,x] /;
FreeQ[{a,b},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && LtQ[p,-1] && GtQ[j*p+1,n-j]
```

2: $\int \left(a\,x^j + b\,x^n\right)^p \,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land p < -1$

Derivation: Generalized binomial recurrence 2b with $m = 0$

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land p < -1$, then

$$\int \left(a\,x^j + b\,x^n\right)^p \,dx \longrightarrow -\frac{\left(a\,x^j + b\,x^n\right)^{p+1}}{a\,(n-j)\,(p+1)\,x^{j-1}} + \frac{n\,p + n - j + 1}{a\,(n-j)\,(p+1)} \int \frac{\left(a\,x^j + b\,x^n\right)^{p+1}}{x^j} \,dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  -(a*x^j+b*x^n)^(p+1)/(a*(n-j)*(p+1)*x^(j-1)) +
  (n*p+n-j+1)/(a*(n-j)*(p+1))*Int[(a*x^j+b*x^n)^(p+1)/x^j,x] /;
FreeQ[{a,b},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && LtQ[p,-1]
```

5. $\int \left(a x^j + b x^n\right)^p dx$ when $p + \frac{1}{2} \in \mathbb{Z} \land j \neq n \land j p + 1 == 0$

1: $\int \left(a x^j + b x^n\right)^p dx$ when $p + \frac{1}{2} \in \mathbb{Z}^+ \land j \neq n \land j p + 1 == 0$

Derivation: Generalized binomial recurrence 1b

Rule: If $p + \frac{1}{2} \in \mathbb{Z}^+ \land j \neq n \land j p + 1 == 0$, then

$$\int \left(a x^j + b x^n\right)^p dx \rightarrow \frac{x \left(a x^j + b x^n\right)^p}{p (n - j)} + a \int x^j \left(a x^j + b x^n\right)^{p-1} dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  x*(a*x^j+b*x^n)^p/(p*(n-j)) + a*Int[x^j*(a*x^j+b*x^n)^(p-1),x] /;
FreeQ[{a,b,j,n},x] && IGtQ[p+1/2,0] && NeQ[n,j] && EqQ[Simplify[j*p+1],0]
```

2. $\int \left(a x^j + b x^n\right)^p dx$ when $p - \frac{1}{2} \in \mathbb{Z}^- \land j \neq n \land j p + 1 == 0$

1: $\int \frac{1}{\sqrt{a x^2 + b x^n}} dx$ when $n \neq 2$

Reference: G&R 2.261.1, CRC 237a, A&S 3.3.33

Reference: CRC 238

Derivation: Integration by substitution

Basis: If $n \neq 2$, then $\frac{1}{\sqrt{a x^2+b x^n}} == \frac{2}{2-n} \text{Subst}\left[\frac{1}{1-a x^2}, x, \frac{x}{\sqrt{a x^2+b x^n}}\right] \partial_x \frac{x}{\sqrt{a x^2+b x^n}}$

Rule: If $n \neq 2$, then

$$\int \frac{1}{\sqrt{a\,x^2+b\,x^n}}\,dx \longrightarrow \frac{2}{2-n}\,\text{Subst}\left[\int \frac{1}{1-a\,x^2}\,dx,\,x,\,\frac{x}{\sqrt{a\,x^2+b\,x^n}}\right]$$

Program code:

```
Int[1/Sqrt[a_.*x_^2+b_.*x_^n_.],x_Symbol] :=
  2/(2-n)*Subst[Int[1/(1-a*x^2),x],x,x/Sqrt[a*x^2+b*x^n]] /;
FreeQ[{a,b,n},x] && NeQ[n,2]
```

2: $\int (a\,x^j+b\,x^n)^p\,dx$ when $p+\frac{1}{2}\in\mathbb{Z}^- \land j\neq n \land j\,p+1==0$

Derivation: Generalized binomial recurrence 2b

Rule: If $p+\frac{1}{2}\in\mathbb{Z}^- \land j\neq n \land j\,p+1==0$, then

$$\int (a\,x^j+b\,x^n)^p\,dx \longrightarrow -\frac{(a\,x^j+b\,x^n)^{p+1}}{a\,(n-j)\,(p+1)\,x^{j-1}}+\frac{n\,p+n-j+1}{a\,(n-j)\,(p+1)}\int \frac{(a\,x^j+b\,x^n)^{p+1}}{x^j}\,dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  -(a*x^j+b*x^n)^(p+1)/(a*(n-j)*(p+1)*x^(j-1)) +
  (n*p+n-j+1)/(a*(n-j)*(p+1))*Int[(a*x^j+b*x^n)^(p+1)/x^j,x] /;
FreeQ[{a,b,j,n},x] && ILtQ[p+1/2,0] && NeQ[n,j] && EqQ[Simplify[j*p+1],0]
```

6: $\int \frac{1}{\sqrt{a x^j + b x^n}} \, dx$ when $2 (n-1) < j < n$

Derivation: Generalized binomial recurrence 3a with $m = 0$ and $p = -\frac{1}{2}$

Rule: If $2 (n-1) < j < n$, then

$$\int \frac{1}{\sqrt{a x^j + b x^n}} \, dx \to -\frac{2 \sqrt{a x^j + b x^n}}{b (n-2) x^{n-1}} - \frac{a (2 n - j - 2)}{b (n-2)} \int \frac{1}{x^{n-j} \sqrt{a x^j + b x^n}} \, dx$$

Program code:

```
Int[1/Sqrt[a_.*x_^j_.+b_.*x_^n_.],x_Symbol] :=
  -2*Sqrt[a*x^j+b*x^n]/(b*(n-2)*x^(n-1)) -
  a*(2*n-j-2)/(b*(n-2))*Int[1/(x^(n-j)*Sqrt[a*x^j+b*x^n]),x] /;
FreeQ[{a,b},x] && LtQ[2*(n-1),j,n]
```

x. $\int (a x^j + b x^n)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n$

1: $\int (a x^j + b x^n)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n \land j p + 1 == 0$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land m + j p + 1 == 0$, then

$$\int (a x^j + b x^n)^p \, dx \longrightarrow \frac{x (a x^j + b x^n)^p}{p (n - j) \left(\frac{a x^j + b x^n}{b x^n}\right)^p} \text{Hypergeometric2F1}\left[-p, -p, 1 - p, -\frac{a}{b x^{n-j}}\right]$$

Program code:

```
(* Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  x*(a*x^j+b*x^n)^p/(p*(n-j)*((a*x^j+b*x^n)/(b*x^n))^p)*Hypergeometric2F1[-p,-p,1-p,-a/(b*x^(n-j))] /;
FreeQ[{a,b,j,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && EqQ[j*p+1,0] *)
```

2: $\int (a x^j + b x^n)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n \land j p + 1 \neq 0$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land j p + 1 \neq 0$, then

$$\int (a x^j + b x^n)^p \, dx \longrightarrow \frac{x (a x^j + b x^n)^p}{(j p + 1) \left(\frac{a x^j + b x^n}{a x^j}\right)^p} \text{Hypergeometric2F1}\left[-p, \frac{j p + 1}{n - j}, \frac{j p + 1}{n - j} + 1, -\frac{b x^{n-j}}{a}\right]$$

Program code:

```
(* Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  x*(a*x^j+b*x^n)^p/((j*p+1)*((a*x^j+b*x^n)/(a*x^j))^p)*
    Hypergeometric2F1[-p,(j*p+1)/(n-j),(j*p+1)/(n-j)+1,-b*x^(n-j)/a] /;
FreeQ[{a,b,j,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && NeQ[j*p+1,0] *)
```

7: $\int \left(a\, x^j + b\, x^n\right)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(a\, x^j+b\, x^n\right)^p}{x^{j\,p} \left(a+b\, x^{n-j}\right)^p} == 0$

Basis: $\frac{\left(a\, x^j+b\, x^n\right)^p}{x^{j\,p} \left(a+b\, x^{n-j}\right)^p} == \frac{\left(a\, x^j+b\, x^n\right)^{\text{FracPart}[p]}}{x^{j\,\text{FracPart}[p]} \left(a+b\, x^{n-j}\right)^{\text{FracPart}[p]}}$

Rule: If $p \notin \mathbb{Z} \land j \neq n$, then

$$\int \left(a\, x^j + b\, x^n\right)^p \, dx \longrightarrow \frac{\left(a\, x^j + b\, x^n\right)^{\text{FracPart}[p]}}{x^{j\,\text{FracPart}[p]} \left(a + b\, x^{n-j}\right)^{\text{FracPart}[p]}} \int x^{j\,p} \left(a + b\, x^{n-j}\right)^p \, dx$$

Program code:

```
Int[(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^n)^FracPart[p]/(x^(j*FracPart[p])*(a+b*x^(n-j))^FracPart[p])*Int[x^(j*p)*(a+b*x^(n-j))^p,x] /;
FreeQ[{a,b,j,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && PosQ[n-j]
```

S: $\int \left(a\, u^j + b\, u^n\right)^p dx$ when $u == c + d\, x$

Derivation: Integration by substitution

Rule: If $u == c + d\, x$, then

$$\int \left(a\, u^j + b\, u^n\right)^p dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int \left(a\, x^j + b\, x^n\right)^p dx,\, x,\, u\right]$$

Program code:

```
Int[(a_.*u_^j_.+b_.*u_^n_.)^p_,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a*x^j+b*x^n)^p,x],x,u] /;
FreeQ[{a,b,j,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```