Rules for integrands of the form $\text{Trig}[d+e\,x]^m\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p$

1. $\int\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$

   1. $\int\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0$

      1: $\int\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0\land p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c=0$, then $a+b\,z+c\,z^2=\frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $b^2-4\,a\,c=0\land p\in\mathbb{Z}$, then

$$\int\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{1}{4^p\,c^p}\int\left(b+2\,c\,\text{Tan}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[(a_+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[(b+2*c*Tan[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[(a_+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[(b+2*c*Cot[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

      2: $\int\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0\land p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c=0$, then $\partial_x\frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{(b+2\,c\,F[x])^{2\,p}}=0$

Rule: If $b^2-4\,a\,c=0\land p\notin\mathbb{Z}$, then

$$\int \left(a + b \operatorname{Tan}[d + e\,x]^n + c \operatorname{Tan}[d + e\,x]^{2\,n}\right)^p \, dx \;\longrightarrow\; \frac{\left(a + b \operatorname{Tan}[d + e\,x]^n + c \operatorname{Tan}[d + e\,x]^{2\,n}\right)^p}{\left(b + 2\,c \operatorname{Tan}[d + e\,x]^n\right)^{2\,p}} \int \left(b + 2\,c \operatorname{Tan}[d + e\,x]^n\right)^{2\,p} dx$$

Program code:

```
Int[(a_+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Tan[d+e*x]^n+c*Tan[d+e*x]^(2*n))^p/(b+2*c*Tan[d+e*x]^n)^(2*p)*Int[(b+2*c*Tan[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[(a_+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cot[d+e*x]^n+c*Cot[d+e*x]^(2*n))^p/(b+2*c*Cot[d+e*x]^n)^(2*p)*Int[(b+2*c*Cot[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int \left(a + b \operatorname{Tan}[d + e x]^n + c \operatorname{Tan}[d + e x]^{2n}\right)^p \, dx$ when $b^2 - 4 a c \neq 0$

1: $\int \frac{1}{a + b \operatorname{Tan}[d + e x]^n + c \operatorname{Tan}[d + e x]^{2n}} \, dx$ when $b^2 - 4 a c \neq 0$

Derivation: Algebraic expansion

Basis: If $q = \sqrt{b^2 - 4 a c}$, then $\frac{1}{a+b z+c z^2} == \frac{2 c}{q (b-q+2 c z)} - \frac{2 c}{q (b+q+2 c z)}$

Rule: If $b^2 - 4 a c \neq 0$, let $q = \sqrt{b^2 - 4 a c}$, then

$$\int \frac{1}{a + b \operatorname{Tan}[d + e x]^n + c \operatorname{Tan}[d + e x]^{2n}} \, dx \to \frac{2 c}{q} \int \frac{1}{b - q + 2 c \operatorname{Tan}[d + e x]^n} \, dx - \frac{2 c}{q} \int \frac{1}{b + q + 2 c \operatorname{Tan}[d + e x]^n} \, dx$$

Program code:

```
Int[1/(a_.+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/(b-q+2*c*Tan[d+e*x]^n),x] -
  2*c/q*Int[1/(b+q+2*c*Tan[d+e*x]^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

```
Int[1/(a_.+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/(b-q+2*c*Cot[d+e*x]^n),x] -
  2*c/q*Int[1/(b+q+2*c*Cot[d+e*x]^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

2. $\int \mathrm{Sin}[d+e\,x]^m\,\left(a+b\,\left(f\,\mathrm{Tan}[d+e\,x]\right)^n+c\,\left(f\,\mathrm{Tan}[d+e\,x]\right)^{2\,n}\right)^p\,dx$

1: $\int \mathrm{Sin}[d+e\,x]^m\,\left(a+b\,\left(f\,\mathrm{Tan}[d+e\,x]\right)^n+c\,\left(f\,\mathrm{Tan}[d+e\,x]\right)^{2\,n}\right)^p\,dx$ when $\frac{m}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Sin}[z]^2 == \frac{\mathrm{Tan}[z]^2}{1+\mathrm{Tan}[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\mathrm{Sin}[d+e\,x]^m\,F[f\,\mathrm{Tan}[d+e\,x]] == \frac{f}{e}\,\mathrm{Subst}\left[\frac{x^m\,F[x]}{\left(f^2+x^2\right)^{\frac{m}{2}+1}},\,x,\,f\,\mathrm{Tan}[d+e\,x]\right]\,\partial_x\,(f\,\mathrm{Tan}[d+e\,x])$$

Rule: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Sin}[d+e\,x]^m\,\left(a+b\,\left(f\,\mathrm{Tan}[d+e\,x]\right)^n+c\,\left(f\,\mathrm{Tan}[d+e\,x]\right)^{2\,n}\right)^p\,dx \longrightarrow \frac{f}{e}\,\mathrm{Subst}\left[\int \frac{x^m\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(f^2+x^2\right)^{\frac{m}{2}+1}}\,dx,\,x,\,f\,\mathrm{Tan}[d+e\,x]\right]$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_*(a_.+b_.*(f_.*tan[d_.+e_.*x_])^n_.+c_.*(f_.*tan[d_.+e_.*x_])^n2_.)^p_,x_Symbol] :=
  f/e*Subst[Int[x^m*(a+b*x^n+c*x^(2*n))^p/(f^2+x^2)^(m/2+1),x],x,f*Tan[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[n2,2*n] && IntegerQ[m/2]
```

```
Int[cos[d_.+e_.*x_]^m_*(a_.+b_.*(f_.*cot[d_.+e_.*x_])^n_.+c_.*(f_.*cot[d_.+e_.*x_])^n2_.)^p_,x_Symbol] :=
  -f/e*Subst[Int[x^m*(a+b*x^n+c*x^(2*n))^p/(f^2+x^2)^(m/2+1),x],x,f*Cot[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[n2,2*n] && IntegerQ[m/2]
```

2: $\int \sin[d+e\,x]^m \left(a+b\,\tan[d+e\,x]^n+c\,\tan[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\tan[z]^2 == \frac{1-\cos[z]^2}{\cos[z]^2}$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}\,\land\,\frac{n}{2}\in\mathbb{Z}$, then

$$\sin[d+e\,x]^m\,F[\tan[d+e\,x]^n] == -\frac{1}{d}\,\text{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}}\,F\left[\frac{\left(1-x^2\right)^{\frac{n}{2}}}{x^n}\right],\,x,\,\cos[d+e\,x]\right]\,\partial_x\cos[d+e\,x]$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$, then

$$\int \sin[d+e\,x]^m \left(a+b\,\tan[d+e\,x]^n+c\,\tan[d+e\,x]^{2\,n}\right)^p\,dx \,\rightarrow\, -\frac{1}{d}\,\text{Subst}\left[\int\frac{\left(1-x^2\right)^{\frac{m-1}{2}}\left(a\,x^{2\,n}+b\,x^n\left(1-x^2\right)^{n/2}+c\left(1-x^2\right)^n\right)^p}{x^{2\,n\,p}}\,dx,\,x,\,\cos[d+e\,x]\right]$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Cos[d+e*x],x]},
  -g/e*Subst[Int[(1-g^2*x^2)^((m-1)/2)*ExpandToSum[a*(g*x)^(2*n)+b*(g*x)^n*(1-g^2*x^2)^(n/2)+c*(1-g^2*x^2)^n,x]^p/(g*x)^(2*n*p),x],x
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && IntegerQ[p]
```

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*cot[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Sin[d+e*x],x]},
  g/e*Subst[Int[(1-g^2*x^2)^((m-1)/2)*ExpandToSum[a*(g*x)^(2*n)+b*(g*x)^n*(1-g^2*x^2)^(n/2)+c*(1-g^2*x^2)^n,x]^p/(g*x)^(2*n*p),x],x,
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && IntegerQ[p]
```

3. $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx$

1: $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Cos}[z]^2 == \frac{1}{1+\text{Tan}[z]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\text{Cos}[d+e\,x]^m\,F[f\,\text{Tan}[d+e\,x]] == \frac{f^{m+1}}{e}\,\text{Subst}\left[\frac{F[x]}{\left(f^2+x^2\right)^{\frac{m}{2}+1}},\ x,\ f\,\text{Tan}[d+e\,x]\right]\partial_x\left(f\,\text{Tan}[d+e\,x]\right)$$

Rule: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx \rightarrow \frac{f^{m+1}}{e}\,\text{Subst}\left[\int \frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(f^2+x^2\right)^{\frac{m}{2}+1}}\,dx,\ x,\ f\,\text{Tan}[d+e\,x]\right]$$

Program code:

```
Int[cos[d_.+e_.*x_]^m_*(a_.+b_.*(f_.*tan[d_.+e_.*x_])^n_.+c_.*(f_.*tan[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  f^(m+1)/e*Subst[Int[(a+b*x^n+c*x^(2*n))^p/(f^2+x^2)^(m/2+1),x],x,f*Tan[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[n2,2*n] && IntegerQ[m/2]
```

```
Int[sin[d_.+e_.*x_]^m_*(a_.+b_.*(f_.*cot[d_.+e_.*x_])^n_.+c_.*(f_.*cot[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  -f^(m+1)/e*Subst[Int[(a+b*x^n+c*x^(2*n))^p/(f^2+x^2)^(m/2+1),x],x,f*Cot[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[n2,2*n] && IntegerQ[m/2]
```

2: $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\in\mathbb{Z}\,\wedge\,\frac{n}{2}\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Tan}[z]^2 == \frac{\text{Sin}[z]^2}{1-\text{Sin}[z]^2}$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}\,\wedge\,\frac{n}{2}\in\mathbb{Z}$, then

$$\text{Cos}[d+e\,x]^m\,F\left[\text{Tan}[d+e\,x]^n\right] == \frac{1}{e}\,\text{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}}\,F\left[\frac{x^n}{\left(1-x^2\right)^{\frac{n}{2}}}\right],\,x,\,\text{Sin}[d+e\,x]\right]\,\partial_x\,\text{Sin}[d+e\,x]$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\,\wedge\,\frac{n}{2}\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}$, then

$$\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx \to \frac{1}{e}\,\text{Subst}\left[\int \left(1-x^2\right)^{(m-2\,n\,p-1)/2}\left(c\,x^{2\,n}+b\,x^n\left(1-x^2\right)^{n/2}+a\left(1-x^2\right)^n\right)^p dx,\,x,\,\text{Sin}[d+e\,x]\right]$$

Program code:

```
Int[cos[d_.+e_.*x_]^m_*(a_.+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Sin[d+e*x],x]},
  g/e*Subst[Int[(1-g^2*x^2)^((m-2*n*p-1)/2)*ExpandToSum[c*x^(2*n)+b*x^n*(1-x^2)^(n/2)+a*(1-x^2)^n,x]^p,x],x,Sin[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && IntegerQ[p]
```

```
Int[sin[d_.+e_.*x_]^m_*(a_.+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Cos[d+e*x],x]},
  -g/e*Subst[Int[(1-g^2*x^2)^((m-2*n*p-1)/2)*ExpandToSum[c*x^(2*n)+b*x^n*(1-x^2)^(n/2)+a*(1-x^2)^n,x]^p,x],x,Cos[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && IntegerQ[p]
```

4. $\int \mathrm{Tan}[d+e\,x]^m\,(a+b\,\mathrm{Tan}[d+e\,x]^n+c\,\mathrm{Tan}[d+e\,x]^{2\,n})^p\,dx$

1. $\int \mathrm{Tan}[d+e\,x]^m\,(a+b\,\mathrm{Tan}[d+e\,x]^n+c\,\mathrm{Tan}[d+e\,x]^{2\,n})^p\,dx$ when $b^2-4\,a\,c==0$

1: $\int \mathrm{Tan}[d+e\,x]^m\,(a+b\,\mathrm{Tan}[d+e\,x]^n+c\,\mathrm{Tan}[d+e\,x]^{2\,n})^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$, then

$$\int \mathrm{Tan}[d+e\,x]^m\,(a+b\,\mathrm{Tan}[d+e\,x]^n+c\,\mathrm{Tan}[d+e\,x]^{2\,n})^p\,dx\,\rightarrow\,\frac{1}{4^p\,c^p}\int \mathrm{Tan}[d+e\,x]^m\,(b+2\,c\,\mathrm{Tan}[d+e\,x]^n)^{2\,p}\,dx$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  1/(4^p*c^p)*Int[Tan[d+e*x]^m*(b+2*c*Tan[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  1/(4^p*c^p)*Int[Cot[d+e*x]^m*(b+2*c*Cot[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \text{Tan}[d+e\,x]^m\,\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x\frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{(b+2\,c\,F[x])^{2\,p}}==0$

Rule: If $b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$, then

$$\int \text{Tan}[d+e\,x]^m\,\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p}{\left(b+2\,c\,\text{Tan}[d+e\,x]^n\right)^{2\,p}}\int \text{Tan}[d+e\,x]^m\,\left(b+2\,c\,\text{Tan}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Tan[d+e*x]^n+c*Tan[d+e*x]^(2*n))^p/(b+2*c*Tan[d+e*x]^n)^(2*p)*Int[Tan[d+e*x]^m*(b+2*c*Tan[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cot[d+e*x]^n+c*Cot[d+e*x]^(2*n))^p/(b+2*c*Cot[d+e*x]^n)^(2*p)*Int[Cot[d+e*x]^m*(b+2*c*Cot[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2: $\int \mathrm{Tan}[d+e\,x]^m\,\left(a+b\,(f\,\mathrm{Tan}[d+e\,x])^n+c\,(f\,\mathrm{Tan}[d+e\,x])^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Integration by substitution

Basis: $\mathrm{Tan}[d+e\,x]^m\,F[f\,\mathrm{Tan}[d+e\,x]] = \frac{f}{e}\,\mathrm{Subst}\left[\left(\frac{x}{f}\right)^m \frac{F[x]}{f^2+x^2},\,x,\,f\,\mathrm{Tan}[d+e\,x]\right]\,\partial_x\,(f\,\mathrm{Tan}[d+e\,x])$

Rule: If $b^2-4\,a\,c\neq 0$, then

$$\int \mathrm{Tan}[d+e\,x]^m\,\left(a+b\,(f\,\mathrm{Tan}[d+e\,x])^n+c\,(f\,\mathrm{Tan}[d+e\,x])^{2\,n}\right)^p\,dx \rightarrow \frac{f}{e}\,\mathrm{Subst}\left[\int \left(\frac{x}{f}\right)^m \frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{f^2+x^2}\,dx,\,x,\,f\,\mathrm{Tan}[d+e\,x]\right]$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*(f_.*tan[d_.+e_.*x_])^n_.+c_.*(f_.*tan[d_.+e_.*x_])^n2_.)^p_,x_Symbol] :=
  f/e*Subst[Int[(x/f)^m*(a+b*x^n+c*x^(2*n))^p/(f^2+x^2),x],x,f*Tan[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*(f_.*cot[d_.+e_.*x_])^n_.+c_.*(f_.*cot[d_.+e_.*x_])^n2_.)^p_,x_Symbol] :=
  -f/e*Subst[Int[(x/f)^m*(a+b*x^n+c*x^(2*n))^p/(f^2+x^2),x],x,f*Cot[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

5. $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$

1. $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0$

1: $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$, then

$$\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p\,dx \to \frac{1}{4^p\,c^p}\int \text{Cot}[d+e\,x]^m \left(b+2\,c\,\text{Tan}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Cot[d+e*x]^m*(b+2*c*Tan[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Tan[d+e*x]^m*(b+2*c*Cot[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx$ when $b^2-4\,a\,c=0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c=0$, then $\partial_x \frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{\left(b+2\,c\,F[x]\right)^{2\,p}}=0$

Rule: If $b^2-4\,a\,c=0\,\wedge\,p\notin\mathbb{Z}$, then

$$\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx \to \frac{\left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p}{\left(b+2\,c\,\text{Tan}[d+e\,x]^n\right)^{2\,p}} \int \text{Cot}[d+e\,x]^m \left(b+2\,c\,\text{Tan}[d+e\,x]^n\right)^{2\,p} dx$$

Program code:

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*tan[d_.+e_.*x_]^n_.+c_.*tan[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Tan[d+e*x]^n+c*Tan[d+e*x]^(2*n))^p/(b+2*c*Tan[d+e*x]^n)^(2*p)*Int[Cot[d+e*x]^m*(b+2*c*Tan[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*cot[d_.+e_.*x_]^n_.+c_.*cot[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cot[d+e*x]^n+c*Cot[d+e*x]^(2*n))^p/(b+2*c*Cot[d+e*x]^n)^(2*p)*Int[Tan[d+e*x]^m*(b+2*c*Cot[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2: $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Tan}[d+e\,x]^n+c\,\text{Tan}[d+e\,x]^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Tan}[z]^2 = \frac{1}{\text{Cot}[z]^2}$

Basis: $\text{Cot}[d+e\,x]^m\,F\left[\text{Tan}[d+e\,x]^2\right] = -\frac{1}{e}\,\text{Subst}\left[\frac{x^m\,F\left[\frac{1}{x^2}\right]}{1+x^2},\,x,\,\text{Cot}[d+e\,x]\right]\partial_x \text{Cot}[d+e\,x]$

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Cot}[d+e\,x]^m\,\left(a+b\,\mathrm{Tan}[d+e\,x]^n+c\,\mathrm{Tan}[d+e\,x]^{2\,n}\right)^p\,dx\;\longrightarrow\;-\frac{1}{e}\,\mathrm{Subst}\left[\int\frac{x^{m-2\,n\,p}\,\left(c+b\,x^n+a\,x^{2\,n}\right)^p}{1+x^2}\,dx,\,x,\,\mathrm{Cot}[d+e\,x]\right]$$

Program code:

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*tan[d_.+e_.*x_]^n_+c_.*tan[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Cot[d+e*x],x]},
  g/e*Subst[Int[(g*x)^(m-2*n*p)*(c+b*(g*x)^n+a*(g*x)^(2*n))^p/(1+g^2*x^2),x],x,Cot[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,m,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2]
```

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*cot[d_.+e_.*x_]^n_+c_.*cot[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Tan[d+e*x],x]},
  -g/e*Subst[Int[(g*x)^(m-2*n*p)*(c+b*(g*x)^n+a*(g*x)^(2*n))^p/(1+g^2*x^2),x],x,Tan[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,m,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2]
```

6. $\int (A + B \tan[d + e x]) (a + b \tan[d + e x] + c \tan[d + e x]^2)^n \, dx$

1. $\int (A + B \tan[d + e x]) (a + b \tan[d + e x] + c \tan[d + e x]^2)^n \, dx$ when $b^2 - 4 a c == 0$

1: $\int (A + B \tan[d + e x]) (a + b \tan[d + e x] + c \tan[d + e x]^2)^n \, dx$ when $b^2 - 4 a c == 0 \land n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2 - 4 a c == 0$, then $a + b z + c z^2 == \frac{(b+2 c z)^2}{4 c}$

Rule: If $b^2 - 4 a c == 0 \land n \in \mathbb{Z}$, then

$$\int (A + B \tan[d + e x]) (a + b \tan[d + e x] + c \tan[d + e x]^2)^n \, dx \to \frac{1}{4^n c^n} \int (A + B \tan[d + e x]) (b + 2 c \tan[d + e x])^{2 n} \, dx$$

Program code:

```
Int[(A_+B_.*tan[d_.+e_.*x_])*(a_+b_.*tan[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  1/(4^n*c^n)*Int[(A+B*Tan[d+e*x])*(b+2*c*Tan[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && IntegerQ[n]
```

```
Int[(A_+B_.*cot[d_.+e_.*x_])*(a_+b_.*cot[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  1/(4^n*c^n)*Int[(A+B*Cot[d+e*x])*(b+2*c*Cot[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && IntegerQ[n]
```

2: $\int (A + B \operatorname{Tan}[d + e x]) \left(a + b \operatorname{Tan}[d + e x] + c \operatorname{Tan}[d + e x]^2\right)^n dx$ when $b^2 - 4 a c == 0 \land n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4 a c == 0$, then $\partial_x \frac{\left(a+b F[x]+c F[x]^2\right)^n}{(b+2 c F[x])^{2 n}} == 0$

Rule: If $b^2 - 4 a c == 0 \land n \notin \mathbb{Z}$, then

$$\int (A + B \operatorname{Tan}[d + e x]) \left(a + b \operatorname{Tan}[d + e x] + c \operatorname{Tan}[d + e x]^2\right)^n dx \to \frac{\left(a + b \operatorname{Tan}[d + e x] + c \operatorname{Tan}[d + e x]^2\right)^n}{(b + 2 c \operatorname{Tan}[d + e x])^{2 n}} \int (A + B \operatorname{Tan}[d + e x]) (b + 2 c \operatorname{Tan}[d + e x])^{2 n} dx$$

Program code:

```
Int[(A_+B_.*tan[d_.+e_.*x_])*(a_+b_.*tan[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  (a+b*Tan[d+e*x]+c*Tan[d+e*x]^2)^n/(b+2*c*Tan[d+e*x])^(2*n)*Int[(A+B*Tan[d+e*x])*(b+2*c*Tan[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[n]]
```

```
Int[(A_+B_.*cot[d_.+e_.*x_])*(a_+b_.*cot[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  (a+b*Cot[d+e*x]+c*Cot[d+e*x]^2)^n/(b+2*c*Cot[d+e*x])^(2*n)*Int[(A+B*Cot[d+e*x])*(b+2*c*Cot[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[n]]
```

2. $\int (A+B\,\text{Tan}[d+e\,x])\,(a+b\,\text{Tan}[d+e\,x]+c\,\text{Tan}[d+e\,x]^2)^n\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \frac{A+B\,\text{Tan}[d+e\,x]}{a+b\,\text{Tan}[d+e\,x]+c\,\text{Tan}[d+e\,x]^2}\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Algebraic expansion

Basis: If $q=\sqrt{b^2-4\,a\,c}$, then $\frac{A+B\,z}{a+b\,z+c\,z^2}==\left(B+\frac{b\,B-2\,A\,c}{q}\right)\frac{1}{b+q+2\,c\,z}+\left(B-\frac{b\,B-2\,A\,c}{q}\right)\frac{1}{b-q+2\,c\,z}$

Rule: If $b^2-4\,a\,c\neq 0$, let $q=\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{A+B\,\text{Tan}[d+e\,x]}{a+b\,\text{Tan}[d+e\,x]+c\,\text{Tan}[d+e\,x]^2}\,dx\to\left(B+\frac{b\,B-2\,A\,c}{q}\right)\int\frac{1}{b+q+2\,c\,\text{Tan}[d+e\,x]}\,dx+\left(B-\frac{b\,B-2\,A\,c}{q}\right)\int\frac{1}{b-q+2\,c\,\text{Tan}[d+e\,x]}\,dx$$

Program code:

```
Int[(A_+B_.*tan[d_.+e_.*x_])/(a_.+b_.*tan[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_]^2),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  (B+(b*B-2*A*c)/q)*Int[1/Simp[b+q+2*c*Tan[d+e*x],x],x] +
  (B-(b*B-2*A*c)/q)*Int[1/Simp[b-q+2*c*Tan[d+e*x],x],x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(A_+B_.*cot[d_.+e_.*x_])/(a_.+b_.*cot[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_]^2),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  (B+(b*B-2*A*c)/q)*Int[1/Simp[b+q+2*c*Cot[d+e*x],x],x] +
  (B-(b*B-2*A*c)/q)*Int[1/Simp[b-q+2*c*Cot[d+e*x],x],x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0]
```

2: $\int (A + B\,\mathrm{Tan}[d + e\,x])\,(a + b\,\mathrm{Tan}[d + e\,x] + c\,\mathrm{Tan}[d + e\,x]^2)^n\,dx$ when $b^2 - 4\,a\,c \neq 0\,\land\,n \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $b^2 - 4\,a\,c \neq 0\,\land\,n \in \mathbb{Z}$

$$\int (A + B\,\mathrm{Tan}[d + e\,x])\,(a + b\,\mathrm{Tan}[d + e\,x] + c\,\mathrm{Tan}[d + e\,x]^2)^n\,dx \longrightarrow \int \mathrm{ExpandTrig}[(A + B\,\mathrm{Tan}[d + e\,x])\,(a + b\,\mathrm{Tan}[d + e\,x] + c\,\mathrm{Tan}[d + e\,x]^2)^n,\,x]\,dx$$

Program code:

```
Int[(A_+B_.*tan[d_.+e_.*x_])*(a_.+b_.*tan[d_.+e_.*x_]+c_.*tan[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  Int[ExpandTrig[(A+B*tan[d+e*x])*(a+b*tan[d+e*x]+c*tan[d+e*x]^2)^n,x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && IntegerQ[n]
```

```
Int[(A_+B_.*cot[d_.+e_.*x_])*(a_.+b_.*cot[d_.+e_.*x_]+c_.*cot[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  Int[ExpandTrig[(A+B*cot[d+e*x])*(a+b*cot[d+e*x]+c*cot[d+e*x]^2)^n,x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && IntegerQ[n]
```