?: $\int u\, P[x]^p\, Q[x]^q\, dx$ when $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^- \land \text{PolyGCD}[P[x], Q[x], x] \neq 1$

Derivation: Algebraic simplification

Rule: If $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^-$, let $gcd = \text{PolyGCD}[P[x], Q[x], x]$, if $gcd \neq 1$, then

$$\int u\, P[x]^p\, Q[x]^q\, dx \to \int u\, gcd^{p+q}\, \text{PolynomialQuotient}[P[x], gcd, x]^p\, \text{PolynomialQuotient}[Q[x], gcd, x]^q\, dx$$

Program code:

```
Int[u_.*P_^p_*Q_^q_,x_Symbol] :=
  Module[{gcd=PolyGCD[P,Q,x]},
  Int[u*gcd^(p+q)*PolynomialQuotient[P,gcd,x]^p*PolynomialQuotient[Q,gcd,x]^q,x] /;
 NeQ[gcd,1]] /;
IGtQ[p,0] && ILtQ[q,0] && PolyQ[P,x] && PolyQ[Q,x]
```

```
Int[u_.*P_*Q_^q_,x_Symbol] :=
  Module[{gcd=PolyGCD[P,Q,x]},
  Int[u*gcd^(q+1)*PolynomialQuotient[P,gcd,x]*PolynomialQuotient[Q,gcd,x]^q,x] /;
 NeQ[gcd,1]] /;
ILtQ[q,0] && PolyQ[P,x] && PolyQ[Q,x]
```

Rules for integrands of the form $P[x]^p$

0: $\int u\,P[x]^p\,dx$ when $p \notin \mathbb{Z} \land P[x] == x^m\,Q[x]$

Derivation: Piecewise constant extraction

Basis: If $P[x] == x^m\,Q[x]$, then $\partial_x \frac{P[x]^p}{x^{m\,p}\,Q[x]^p} == 0$

Rule: If $p \notin \mathbb{Z} \land P[x] == x^m\,Q[x]$, then

$$\int u\,P[x]^p\,dx \longrightarrow \frac{P[x]^{\text{FracPart}[p]}}{x^{m\,\text{FracPart}[p]}\,Q[x]^{\text{FracPart}[p]}}\int u\,x^{m\,p}\,Q[x]^p\,dx$$

Program code:

```
Int[u_.*P_^p_.,x_Symbol] :=
  With[{m=MinimumMonomialExponent[P,x]},
  P^FracPart[p]/(x^(m*FracPart[p])*Distrib[1/x^m,P]^FracPart[p])*Int[u*x^(m*p)*Distrib[1/x^m,P]^p,x]] /;
FreeQ[p,x] && Not[IntegerQ[p]] && SumQ[P] && EveryQ[Function[BinomialQ[#,x]],P] && Not[PolyQ[P,x,2]]
```

1. $\int P[x]^p \, dx$ when $P[x] == P1[x] \, P2[x] \cdots$

   1: $\int P[x^2]^p \, dx$ when $p \in \mathbb{Z}^- \land P[x] == P1[x] \, P2[x] \cdots$

Derivation: Algebraic simplification

Note: This rule assumes host CAS distributes integer powers over products.

Rule: If $p \in \mathbb{Z}^- \land P[x] == P1[x] \, P2[x] \cdots$, then

$$\int P[x^2]^p \, dx \to \int \text{ExpandIntegrand}\left[P1[x^2]^p \, P2[x^2]^p \cdots, x\right] dx$$

Program code:

```
Int[P_^p_,x_Symbol] :=
  With[{u=Factor[ReplaceAll[P,x→Sqrt[x]]]},
  Int[ExpandIntegrand[ReplaceAll[u,x→x^2]^p,x],x] /;
 Not[SumQ[NonfreeFactors[u,x]]]] /;
PolyQ[P,x^2] && ILtQ[p,0]
```

2: $\int P[x]^p \, dx$ when $p \in \mathbb{Z}^- \land P[x] == P1[x] \, P2[x] \cdots$

Derivation: Algebraic simplification

Note: This rule assumes host CAS distributes integer powers over products.

Rule: If $p \in \mathbb{Z}^- \land P[x] == P1[x] \, P2[x] \cdots$, then

$$\int P[x]^p \, dx \longrightarrow \int P1[x]^p \, P2[x]^p \cdots dx$$

Program code:

```
Int[P_^p_,x_Symbol] :=
  With[{u=Factor[P]},
  Int[ExpandIntegrand[u^p,x],x] /;
 Not[SumQ[NonfreeFactors[u,x]]]] /;
PolyQ[P,x] && ILtQ[p,0]
```

2: $\int P[x]^p \, dx$ when $p \in \mathbb{Z} \land P[x] == P1[x] \, P2[x] \cdots$

Derivation: Algebraic simplification

Note: This rule assumes host CAS distributes integer powers over products.

Rule: If $p \in \mathbb{Z} \land P[x] == P1[x] \, P2[x] \cdots$, then

$$\int P[x]^p \, dx \longrightarrow \int P1[x]^p \, P2[x]^p \cdots \, dx$$

Program code:

```
Int[P_^p_,x_Symbol] :=
  With[{u=Factor[P]},
  Int[u^p,x] /;
 Not[SumQ[NonfreeFactors[u,x]]]] /;
PolyQ[P,x] && IntegerQ[p]
```

x: $\int P_n[x]^p \, dx$ when $P_n[x] == Q_{n1}[x]^q R_{n2}[x]^r \cdots \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $P_n[x] == Q_{n1}[x]^q R_{n2}[x]^r \cdots$, then $\partial_x \frac{P_n[x]^p}{Q_{n1}[x]^{p\,q} R_{n2}[x]^{p\,r} \cdots} == 0$

Rule: If $P_n[x] == Q_{n1}[x]^q R_{n2}[x]^r \cdots \land p \notin \mathbb{Z}$, then

$$\int P_n[x]^p \, dx \longrightarrow \frac{P_n[x]^p}{Q_{n1}[x]^{p\,q} R_{n2}[x]^{p\,r} \cdots} \int Q_{n1}[x]^{p\,q} R_{n2}[x]^{p\,r} \cdots \, dx$$

Program code:

```
(* Int[Pn_^p_,x_Symbol] :=
  With[{u=Factor[Pn]},
  Pn^p/DistributeDegree[u,p]*Int[DistributeDegree[u,p],x] /;
 Not[SumQ[u]]] /;
PolyQ[Pn,x] && Not[IntegerQ[p]] *)
```

2. $\int P[x]^p \, dx$ when $p \in \mathbb{Z}^+$

1: $\int (a + b x + c x^2 + d x^3)^p \, dx$ when $p \in \mathbb{Z}^+ \land c^2 - 3 b d == 0$

Derivation: Integration by substitution

Basis: If $c^2 - 3 b d == 0$, then $(a + b x + c x^2 + d x^3)^p == \frac{1}{3^p} \text{Subst}\left[\left(\frac{3 a c - b^2}{c} + \frac{c^2 x^3}{b}\right)^p, x, \frac{c}{3 d} + x\right] \partial_x \left(\frac{c}{3 d} + x\right)$

Rule: If $p \in \mathbb{Z}^+ \land c^2 - 3 b d == 0$, then

$$\int (a + b x + c x^2 + d x^3)^p \, dx \longrightarrow \frac{1}{3^p} \text{Subst}\left[\int \left(\frac{3 a c - b^2}{c} + \frac{c^2 x^3}{b}\right)^p dx, x, \frac{c}{3 d} + x\right]$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/3^p*Subst[Int[Simp[(3*a*c-b^2)/c+c^2*x^3/b,x]^p,x],x,c/(3*d)+x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0] && EqQ[c^2-3*b*d,0]
```

2: $\int P[x]^p \, dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int P[x]^p \, dx \longrightarrow \int \text{ExpandToSum}\left[P[x]^p, x\right] dx$$

Program code:

```
Int[P_^p_,x_Symbol] :=
  Int[ExpandToSum[P^p,x],x] /;
PolyQ[P,x] && IGtQ[p,0]
```

3: $\int P[x]^p \, dx$ when $p \in \mathbb{Z} \land P[x] == \left(a + b x + c x^2\right) \left(d + e x + f x^2\right) \cdots$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z} \land P[x] == \left(a + b x + c x^2\right) \left(d + e x + f x^2\right) \cdots$, then

$$\int P[x]^p \, dx \longrightarrow \int \text{ExpandIntegrand}\left[P[x]^p, x\right] dx$$

Program code:

```
Int[P_^p_,x_Symbol] :=
  Int[ExpandIntegrand[P^p,x],x] /;
PolyQ[P,x] && IntegerQ[p] && QuadraticProductQ[Factor[P],x]
```

4. $\int (a + b x + c x^2 + d x^3)^p \, dx$

1. $\int (a + b x + d x^3)^p \, dx$

1. $\int (a + b x + d x^3)^p \, dx$ when $4 b^3 + 27 a^2 d == 0$

1: $\int (a + b x + d x^3)^p \, dx$ when $4 b^3 + 27 a^2 d == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $4 b^3 + 27 a^2 d == 0$, then $a + b x + d x^3 == \frac{1}{3^3 a^2} (3 a - b x) (3 a + 2 b x)^2$

Rule: If $4 b^3 + 27 a^2 d == 0 \land p \in \mathbb{Z}$, then

$$\int (a + b x + d x^3)^p \, dx \longrightarrow \frac{1}{3^{3 p} a^{2 p}} \int (3 a - b x)^p (3 a + 2 b x)^{2 p} \, dx$$

Program code:

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^(3*p)*a^(2*p))*Int[(3*a-b*x)^p*(3*a+2*b*x)^(2*p),x] /;
FreeQ[{a,b,d},x] && EqQ[4*b^3+27*a^2*d,0] && IntegerQ[p]
```

2: $\int (a + b x + d x^3)^p \, dx$ when $4 b^3 + 27 a^2 d == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $4 b^3 + 27 a^2 d == 0$, then $\partial_x \frac{(a+b x+d x^3)^p}{(3 a-b x)^p (3 a+2 b x)^{2 p}} == 0$

Rule: If $4 b^3 + 27 a^2 d == 0 \land p \notin \mathbb{Z}$, then

$$\int \left(a+b\,x+d\,x^3\right)^p dx \;\longrightarrow\; \frac{\left(a+b\,x+d\,x^3\right)^p}{(3\,a-b\,x)^p\,(3\,a+2\,b\,x)^{2\,p}} \int (3\,a-b\,x)^p\,(3\,a+2\,b\,x)^{2\,p}\,dx$$

Program code:

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  (a+b*x+d*x^3)^p/((3*a-b*x)^p*(3*a+2*b*x)^(2*p))*Int[(3*a-b*x)^p*(3*a+2*b*x)^(2*p),x] /;
FreeQ[{a,b,d,p},x] && EqQ[4*b^3+27*a^2*d,0] && Not[IntegerQ[p]]
```

2. $\int \left(a+b\,x+d\,x^3\right)^p dx$ when $4\,b^3+27\,a^2\,d \neq 0$

1: $\int \left(a+b\,x+d\,x^3\right)^p dx$ when $4\,b^3+27\,a^2\,d \neq 0 \land p \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $r \to \left(-9\,a\,d^2+\sqrt{3}\,d\,\sqrt{4\,b^3\,d+27\,a^2\,d^2}\right)^{1/3}$, then $a+b\,x+d\,x^3 == \frac{2\,b^3\,d}{3\,r^3}-\frac{r^3}{18\,d^2}+b\,x+d\,x^3$

Basis:
$$\frac{2\,b^3\,d}{3\,r^3}-\frac{r^3}{18\,d^2}+b\,x+d\,x^3 == \frac{1}{d^2}\left(\frac{18^{1/3}\,b\,d}{3\,r}-\frac{r}{18^{1/3}}+d\,x\right)\left(\frac{b\,d}{3}+\frac{12^{1/3}\,b^2\,d^2}{3\,r^2}+\frac{r^2}{3\times 12^{1/3}}-d\left(\frac{2^{1/3}\,b\,d}{3^{1/3}\,r}-\frac{r}{18^{1/3}}\right)x+d^2\,x^2\right)$$

Rule: If $4\,b^3+27\,a^2\,d \neq 0 \land p \in \mathbb{Z}$, let $r \to \left(-9\,a\,d^2+\sqrt{3}\,d\,\sqrt{4\,b^3\,d+27\,a^2\,d^2}\right)^{1/3}$, then

$$\int \left(a+b\,x+d\,x^3\right)^p dx \;\longrightarrow$$
$$\frac{1}{d^{2\,p}} \int \left(\frac{18^{1/3}\,b\,d}{3\,r}-\frac{r}{18^{1/3}}+d\,x\right)^p \left(\frac{b\,d}{3}+\frac{12^{1/3}\,b^2\,d^2}{3\,r^2}+\frac{r^2}{3\times 12^{1/3}}-d\left(\frac{2^{1/3}\,b\,d}{3^{1/3}\,r}-\frac{r}{18^{1/3}}\right)x+d^2\,x^2\right)^p dx$$

Program code:

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-9*a*d^2+Sqrt[3]*d*Sqrt[4*b^3*d+27*a^2*d^2],3]},
  1/d^(2*p)*Int[Simp[18^(1/3)*b*d/(3*r)-r/18^(1/3)+d*x,x]^p*
    Simp[b*d/3+12^(1/3)*b^2*d^2/(3*r^2)+r^2/(3*12^(1/3))-d*(2^(1/3)*b*d/(3^(1/3)*r)-r/18^(1/3))*x+d^2*x^2,x]^p,x]] /;
FreeQ[{a,b,d},x] && NeQ[4*b^3+27*a^2*d,0] && IntegerQ[p]
```

2: $\int \left(a+b\,x+d\,x^3\right)^p\,dx$ when $4\,b^3+27\,a^2\,d\neq 0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $r\to\left(-9\,a\,d^2+\sqrt{3}\,d\,\sqrt{4\,b^3\,d+27\,a^2\,d^2}\right)^{1/3}$, then

$$\partial_x\left(\left(a+b\,x+d\,x^3\right)^p\Big/\left(\left(\frac{18^{1/3}\,b\,d}{3\,r}-\frac{r}{18^{1/3}}+d\,x\right)^p\left(\frac{b\,d}{3}+\frac{12^{1/3}\,b^2\,d^2}{3\,r^2}+\frac{r^2}{3\times 12^{1/3}}-d\left(\frac{2^{1/3}\,b\,d}{3^{1/3}\,r}-\frac{r}{18^{1/3}}\right)x+d^2\,x^2\right)^p\right)\right)=0$$

Rule: If $4\,b^3+27\,a^2\,d\neq 0\,\land\,p\notin\mathbb{Z}$, let $r\to\left(-9\,a\,d^2+\sqrt{3}\,d\,\sqrt{4\,b^3\,d+27\,a^2\,d^2}\right)^{1/3}$, then

$$\int\left(a+b\,x+d\,x^3\right)^p\,dx\;\longrightarrow$$

$$\left(\left(a+b\,x+d\,x^3\right)^p\Big/\left(\left(\frac{18^{1/3}\,b\,d}{3\,r}-\frac{r}{18^{1/3}}+d\,x\right)^p\left(\frac{b\,d}{3}+\frac{12^{1/3}\,b^2\,d^2}{3\,r^2}+\frac{r^2}{3\times 12^{1/3}}-d\left(\frac{2^{1/3}\,b\,d}{3^{1/3}\,r}-\frac{r}{18^{1/3}}\right)x+d^2\,x^2\right)^p\right)\right)\cdot$$

$$\int\left(\frac{18^{1/3}\,b\,d}{3\,r}-\frac{r}{18^{1/3}}+d\,x\right)^p\left(\frac{b\,d}{3}+\frac{12^{1/3}\,b^2\,d^2}{3\,r^2}+\frac{r^2}{3\times 12^{1/3}}-d\left(\frac{2^{1/3}\,b\,d}{3^{1/3}\,r}-\frac{r}{18^{1/3}}\right)x+d^2\,x^2\right)^p\,dx$$

Program code:

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-9*a*d^2+Sqrt[3]*d*Sqrt[4*b^3*d+27*a^2*d^2],3]},
  (a+b*x+d*x^3)^p/
    (Simp[18^(1/3)*b*d/(3*r)-r/18^(1/3)+d*x,x]^p*
      Simp[b*d/3+12^(1/3)*b^2*d^2/(3*r^2)+r^2/(3*12^(1/3))-d*(2^(1/3)*b*d/(3^(1/3)*r)-r/18^(1/3))*x+d^2*x^2,x]^p)*
    Int[Simp[18^(1/3)*b*d/(3*r)-r/18^(1/3)+d*x,x]^p*
      Simp[b*d/3+12^(1/3)*b^2*d^2/(3*r^2)+r^2/(3*12^(1/3))-d*(2^(1/3)*b*d/(3^(1/3)*r)-r/18^(1/3))*x+d^2*x^2,x]^p,x]] /;
FreeQ[{a,b,d,p},x] && NeQ[4*b^3+27*a^2*d,0] && Not[IntegerQ[p]]
```

2: $\int \left(a + b x + c x^2 + d x^3\right)^p \, dx$

Derivation: Integration by substitution

Rule:

$$\int \left(a + b x + c x^2 + d x^3\right)^p dx \longrightarrow \text{Subst}\left[\int \left(\frac{2 c^3 - 9 b c d + 27 a d^2}{27 d^2} - \frac{\left(c^2 - 3 b d\right) x}{3 d} + d x^3\right)^p dx,\ x,\ x + \frac{c}{3 d}\right]$$

Program code:

```
Int[P3_^p_,x_Symbol] :=
  With[{a=Coeff[P3,x,0],b=Coeff[P3,x,1],c=Coeff[P3,x,2],d=Coeff[P3,x,3]},
  Subst[Int[Simp[(2*c^3-9*b*c*d+27*a*d^2)/(27*d^2)-(c^2-3*b*d)*x/(3*d)+d*x^3,x]^p,x],x,x+c/(3*d)] /;
 NeQ[c,0]] /;
FreeQ[p,x] && PolyQ[P3,x,3]
```

5. $\int (a+b x+c x^2+d x^3+e x^4)^p \, dx$

1: $\int (a+b x+c x^2+d x^3+e x^4)^p \, dx$ when $p \in \mathbb{Z}^- \land a \neq 0 \land c == \frac{b^2}{a} \land d == \frac{b^3}{a^2} \land e == \frac{b^4}{a^3}$

Derivation: Algebraic simplification

Basis: If $a \neq 0 \land c == \frac{b^2}{a} \land d == \frac{b^3}{a^2} \land e == \frac{b^4}{a^3}$, then $a+b x+c x^2+d x^3+e x^4 == \frac{a^5-b^5 x^5}{a^3 (a-b x)}$

Rule: If $p \in \mathbb{Z}^- \land a \neq 0 \land c == \frac{b^2}{a} \land d == \frac{b^3}{a^2} \land e == \frac{b^4}{a^3}$, then

$$\int (a+b x+c x^2+d x^3+e x^4)^p \, dx \to \frac{1}{a^{3p}} \int \text{ExpandIntegrand}\left[\frac{(a-b x)^{-p}}{(a^5-b^5 x^5)^{-p}}, x\right] dx$$

Program code:

```
Int[P4_^p_,x_Symbol] :=
  With[{a=Coeff[P4,x,0],b=Coeff[P4,x,1],c=Coeff[P4,x,2],d=Coeff[P4,x,3],e=Coeff[P4,x,4]},
  1/a^(3*p)*Int[ExpandIntegrand[(a-b*x)^(-p)/(a^5-b^5*x^5)^(-p),x],x] /;
 NeQ[a,0] && EqQ[c,b^2/a] && EqQ[d,b^3/a^2] && EqQ[e,b^4/a^3]] /;
FreeQ[p,x] && PolyQ[P4,x,4] && ILtQ[p,0]
```

2: $\int (a+b x+c x^2+d x^3+e x^4)^p \, dx$ when $b^3-4 a b c+8 a^2 d == 0 \land 2 p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $b^3-4 a b c+8 a^2 d == 0$, then

$$(a+b x+c x^2+d x^3+e x^4)^p == -16 a^2 \, \text{Subst}\Big[\frac{1}{(b-4 a x)^2} \left(\frac{1}{(b-4 a x)^4} a \left(-3 b^4+16 a b^2 c-64 a^2 b d+256 a^3 e-32 a^2 (3 b^2-8 a c) x^2+256 a^4 x^4\right)\right)^p, x, \frac{b}{4 a}+\frac{1}{x}\Big] \partial_x \left(\frac{b}{4 a}+\frac{1}{x}\right)$$

Note: The substitution transforms a dense quartic polynomial into a symmetric quartic trinomial over the 4th power of a linear.

Rule: If $b^3 - 4 a b c + 8 a^2 d == 0 \wedge 2 p \in \mathbb{Z}$, then

$$\int \left(a + b x + c x^2 + d x^3 + e x^4\right)^p dx \longrightarrow$$

$$-16 a^2 \text{Subst}\left[\int \frac{1}{(b - 4 a x)^2} \left(\frac{1}{(b - 4 a x)^4} a \left(-3 b^4 + 16 a b^2 c - 64 a^2 b d + 256 a^3 e - 32 a^2 \left(3 b^2 - 8 a c\right) x^2 + 256 a^4 x^4\right)\right)^p dx,\ x,\ \frac{b}{4 a} + \frac{1}{x}\right]$$

Program code:

```
Int[P4_^p_,x_Symbol] :=
  With[{a=Coeff[P4,x,0],b=Coeff[P4,x,1],c=Coeff[P4,x,2],d=Coeff[P4,x,3],e=Coeff[P4,x,4]},
  -16*a^2*Subst[
    Int[1/(b-4*a*x)^2*(a*(-3*b^4+16*a*b^2*c-64*a^2*b*d+256*a^3*e-32*a^2*(3*b^2-8*a*c)*x^2+256*a^4*x^4)/(b-4*a*x)^4)^p,x],
    x,b/(4*a)+1/x] /;
 NeQ[a,0] && NeQ[b,0] && EqQ[b^3-4*a*b*c+8*a^2*d,0]] /;
FreeQ[p,x] && PolyQ[P4,x,4] && IntegerQ[2*p] && Not[IGtQ[p,0]]
```

6: $\int \left(a + b x^2 + c x^3 + d x^4 + e x^6\right)^p \, dx$ when $p \in \mathbb{Z}^- \land b^2 - 3 a d = 0 \land b^3 - 27 a^2 e = 0$

Algebraic expansion

Basis: If $b^2 - 3 a d = 0 \land b^3 - 27 a^2 e = 0$, then
$a + b x^2 + c x^3 + d x^4 + e x^6 = \frac{1}{27 a^2} \left(3 a + 3 a^{2/3} c^{1/3} x + b x^2\right) \left(3 a - 3 (-1)^{1/3} a^{2/3} c^{1/3} x + b x^2\right) \left(3 a + 3 (-1)^{2/3} a^{2/3} c^{1/3} x + b x^2\right)$

Note: If $\frac{m+1}{2} \in \mathbb{Z}^+$, then $c x^m + \left(a + b x^2\right)^m = \prod_{k=1}^{m} \left(a + (-1)^{k \left(1 - \frac{1}{m}\right)} c^{\frac{1}{m}} x + b x^2\right)$

Rule: If $p \in \mathbb{Z}^- \land b^2 - 3 a d = 0 \land b^3 - 27 a^2 e = 0$, then

$$\int \left(a + b x^2 + c x^3 + d x^4 + e x^6\right)^p \, dx \longrightarrow$$
$$\frac{1}{3^{3 p} a^{2 p}} \int \text{ExpandIntegrand}\left[\left(3 a + 3 a^{2/3} c^{1/3} x + b x^2\right)^p \left(3 a - 3 (-1)^{1/3} a^{2/3} c^{1/3} x + b x^2\right)^p \left(3 a + 3 (-1)^{2/3} a^{2/3} c^{1/3} x + b x^2\right)^p, x\right] dx$$

Program code:

```
Int[Q6_^p_,x_Symbol] :=
  With[{a=Coeff[Q6,x,0],b=Coeff[Q6,x,2],c=Coeff[Q6,x,3],d=Coeff[Q6,x,4],e=Coeff[Q6,x,6]},
  1/(3^(3*p)*a^(2*p))*Int[ExpandIntegrand[
    (3*a+3*Rt[a,3]^2*Rt[c,3]*x+b*x^2)^p*
    (3*a-3*(-1)^(1/3)*Rt[a,3]^2*Rt[c,3]*x+b*x^2)^p*
    (3*a+3*(-1)^(2/3)*Rt[a,3]^2*Rt[c,3]*x+b*x^2)^p,x],x] /;
 EqQ[b^2-3*a*d,0] && EqQ[b^3-27*a^2*e,0]] /;
ILtQ[p,0] && PolyQ[Q6,x,6] && EqQ[Coeff[Q6,x,1],0] && EqQ[Coeff[Q6,x,5],0] && RationalFunctionQ[u,x]
```