Rules for integrands of the form $(c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n$

1. $\int (c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx$

1: $\int (c+d\,x)^m\,\mathrm{Tan}[e+f\,x]\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\mathrm{Tan}[z] == i - \frac{2\,i\,e^{2\,i\,z}}{1+e^{2\,i\,z}} == -i + \frac{2\,i\,e^{-2\,i\,z}}{1+e^{-2\,i\,z}}$

Rule: If $m\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,\mathrm{Tan}[e+f\,x]\,dx \to \frac{i\,(c+d\,x)^{m+1}}{d\,(m+1)} - 2\,i\int\frac{(c+d\,x)^m\,e^{2\,i\,(e+f\,x)}}{1+e^{2\,i\,(e+f\,x)}}\,dx$$

$$\int (c+d\,x)^m\,\mathrm{Tan}[e+f\,x]\,dx \to -\frac{i\,(c+d\,x)^{m+1}}{d\,(m+1)} + 2\,i\int\frac{(c+d\,x)^m\,e^{-2\,i\,(e+f\,x)}}{1+e^{-2\,i\,(e+f\,x)}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*tan[e_.+k_.*Pi+f_.*Complex[0,fz_]*x_],x_Symbol] :=
  -I*(c+d*x)^(m+1)/(d*(m+1)) + 2*I*Int[(c+d*x)^m*E^(-2*I*k*Pi)*E^(2*(-I*e+f*fz*x))/(1+E^(-2*I*k*Pi)*E^(2*(-I*e+f*fz*x))),x] /;
FreeQ[{c,d,e,f,fz},x] && IntegerQ[4*k] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*tan[e_.+k_.*Pi+f_.*x_],x_Symbol] :=
  I*(c+d*x)^(m+1)/(d*(m+1)) - 2*I*Int[(c+d*x)^m*E^(2*I*k*Pi)*E^(2*I*(e+f*x))/(1+E^(2*I*k*Pi)*E^(2*I*(e+f*x))),x] /;
FreeQ[{c,d,e,f},x] && IntegerQ[4*k] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*tan[e_.+f_.*Complex[0,fz_]*x_],x_Symbol] :=
  -I*(c+d*x)^(m+1)/(d*(m+1)) + 2*I*Int[(c+d*x)^m*E^(2*(-I*e+f*fz*x))/(1+E^(2*(-I*e+f*fz*x))),x] /;
FreeQ[{c,d,e,f,fz},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*tan[e_.+f_.*x_],x_Symbol] :=
  I*(c+d*x)^(m+1)/(d*(m+1)) - 2*I*Int[(c+d*x)^m*E^(2*I*(e+f*x))/(1+E^(2*I*(e+f*x))),x] /;
FreeQ[{c,d,e,f},x] && IGtQ[m,0]
```

2: $\int (c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $n>1\land m>0$

Note: This rule does not appear in published integral tables.

Rule: If $n>1\land m>0$, then

$$\int (c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx\;\longrightarrow$$
$$\frac{b\,(c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^{n-1}}{f\,(n-1)}-\frac{b\,d\,m}{f\,(n-1)}\int (c+d\,x)^{m-1}\,(b\,\mathrm{Tan}[e+f\,x])^{n-1}\,dx-b^2\int (c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^{n-2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(c+d*x)^m*(b*Tan[e+f*x])^(n-1)/(f*(n-1)) -
  b*d*m/(f*(n-1))*Int[(c+d*x)^(m-1)*(b*Tan[e+f*x])^(n-1),x] -
  b^2*Int[(c+d*x)^m*(b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{b,c,d,e,f},x] && GtQ[n,1] && GtQ[m,0]
```

3: $\int (c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $n<-1\,\land\,m>0$

Note: This rule does not appear in published integral tables.

Rule: If $n<-1\,\land\,m>0$, then

$$\int (c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx \longrightarrow$$
$$\frac{(c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^{n+1}}{b\,f\,(n+1)}-\frac{d\,m}{b\,f\,(n+1)}\int (c+d\,x)^{m-1}\,(b\,\mathrm{Tan}[e+f\,x])^{n+1}\,dx-\frac{1}{b^2}\int (c+d\,x)^m\,(b\,\mathrm{Tan}[e+f\,x])^{n+2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (c+d*x)^m*(b*Tan[e+f*x])^(n+1)/(b*f*(n+1)) -
  d*m/(b*f*(n+1))*Int[(c+d*x)^(m-1)*(b*Tan[e+f*x])^(n+1),x] -
  1/b^2*Int[(c+d*x)^m*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{b,c,d,e,f},x] && LtQ[n,-1] && GtQ[m,0]
```

2: $\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $(m\mid n)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx \to \int (c+d\,x)^m\,\mathrm{ExpandIntegrand}\left[(a+b\,\mathrm{Tan}[e+f\,x])^n,\,x\right]dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,(a+b*Tan[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[m,0] && IGtQ[n,0]
```

3. $\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\land n\in\mathbb{Z}^-$

1. $\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2==0$

1: $\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2==0\land m>0$

Derivation: Algebraic expansion and integration by parts

Basis: If $a^2+b^2==0$, then $\frac{1}{a+b\,\mathrm{Tan}[z]}==\frac{1}{2\,a}+\frac{a\,\mathrm{Sec}[z]^2}{2\,(a+b\,\mathrm{Tan}[z])^2}$

Basis: $\frac{\mathrm{Sec}[e+f\,x]^2}{(a+b\,\mathrm{Tan}[e+f\,x])^2}==-\partial_x\frac{1}{b\,f\,(a+b\,\mathrm{Tan}[e+f\,x])}$

Rule: If $a^2+b^2==0\land m>0$, then

$$\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx \longrightarrow \frac{(c+d\,x)^{m+1}}{2\,a\,d\,(m+1)}+\frac{a}{2}\int \frac{(c+d\,x)^m\,\mathrm{Sec}[e+f\,x]^2}{(a+b\,\mathrm{Tan}[e+f\,x])^2}\,dx$$

$$\longrightarrow \frac{(c+d\,x)^{m+1}}{2\,a\,d\,(m+1)}-\frac{a\,(c+d\,x)^m}{2\,b\,f\,(a+b\,\mathrm{Tan}[e+f\,x])}+\frac{a\,d\,m}{2\,b\,f}\int \frac{(c+d\,x)^{m-1}}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (c+d*x)^(m+1)/(2*a*d*(m+1)) -
  a*(c+d*x)^m/(2*b*f*(a+b*Tan[e+f*x])) +
  a*d*m/(2*b*f)*Int[(c+d*x)^(m-1)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2+b^2,0] && GtQ[m,0]
```

2. $\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2==0\land m<-1$

1: $\int \frac{1}{(c+d\,x)^2\,(a+b\,\mathrm{Tan}[e+f\,x])}\,dx$ when $a^2+b^2==0$

Derivation: Integration by parts and algebraic expansion

Basis: $\frac{1}{(c+d\,x)^2}==-\partial_x\frac{1}{d\,(c+d\,x)}$

Basis: If $a^2+b^2==0$, then $\partial_x\frac{1}{a+b\,\mathrm{Tan}[e+f\,x]}==\frac{f\,\mathrm{Cos}[2\,e+2\,f\,x]}{b}-\frac{f\,\mathrm{Sin}[2\,e+2\,f\,x]}{a}$

Rule: If $a^2+b^2==0$, then

$$\int \frac{1}{(c+d\,x)^2\,(a+b\,\mathrm{Tan}[e+f\,x])}\,dx\to-\frac{1}{d\,(c+d\,x)\,(a+b\,\mathrm{Tan}[e+f\,x])}+\frac{f}{b\,d}\int\frac{\mathrm{Cos}[2\,e+2\,f\,x]}{c+d\,x}\,dx-\frac{f}{a\,d}\int\frac{\mathrm{Sin}[2\,e+2\,f\,x]}{c+d\,x}\,dx$$

Program code:

```
Int[1/((c_.+d_.*x_)^2*(a_+b_.*tan[e_.+f_.*x_])),x_Symbol] :=
  -1/(d*(c+d*x)*(a+b*Tan[e+f*x])) +
  f/(b*d)*Int[Cos[2*e+2*f*x]/(c+d*x),x] -
  f/(a*d)*Int[Sin[2*e+2*f*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2+b^2,0]
```

2: $\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2==0\land m<-1\land m\neq-2$

Derivation: Previous rule inverted

Rule: If $a^2+b^2==0\land m<-1\land m\neq-2$, then

$$\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx \longrightarrow \frac{f\,(c+d\,x)^{m+2}}{b\,d^2\,(m+1)\,(m+2)}+\frac{(c+d\,x)^{m+1}}{d\,(m+1)\,(a+b\,\mathrm{Tan}[e+f\,x])}+\frac{2\,b\,f}{a\,d\,(m+1)}\int \frac{(c+d\,x)^{m+1}}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  f*(c+d*x)^(m+2)/(b*d^2*(m+1)*(m+2)) +
  (c+d*x)^(m+1)/(d*(m+1)*(a+b*Tan[e+f*x])) +
  2*b*f/(a*d*(m+1))*Int[(c+d*x)^(m+1)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2+b^2,0] && LtQ[m,-1] && NeQ[m,-2]
```

x: $\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2==0\,\land\,m<-1$

Derivation: Previous rule inverted

Note: Although this rule unifies the above two rules, it requires an additional step and when $m=-2$ it generates two log terms that cancel out.

Rule: If $a^2+b^2==0\,\land\,m<-1$, then

$$\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx \longrightarrow \frac{(c+d\,x)^{m+1}}{d\,(m+1)\,(a+b\,\mathrm{Tan}[e+f\,x])}+\frac{f}{b\,d\,(m+1)}\int (c+d\,x)^{m+1}\,dx+\frac{2\,b\,f}{a\,d\,(m+1)}\int \frac{(c+d\,x)^{m+1}}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
(* Int[(c_.+d_.*x_)^m_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (c+d*x)^(m+1)/(d*(m+1)*(a+b*Tan[e+f*x])) +
  f/(b*d*(m+1))*Int[(c+d*x)^(m+1),x] +
  2*b*f/(a*d*(m+1))*Int[(c+d*x)^(m+1)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2+b^2,0] && LtQ[m,-1] *)
```

3: $\int \frac{1}{(c+d\,x)\,(a+b\,\mathrm{Tan}[e+f\,x])}\,dx$ when $a^2+b^2==0$

Derivation: Algebraic expansion

Basis: If $a^2+b^2==0$, then $\frac{1}{a+b\,\mathrm{Tan}[z]}==\frac{1}{2\,a}+\frac{\mathrm{Cos}[2\,z]}{2\,a}+\frac{\mathrm{Sin}[2\,z]}{2\,b}$

Rule: If $a^2+b^2==0$, then

$$\int \frac{1}{(c+d\,x)\,(a+b\,\mathrm{Tan}[e+f\,x])}\,dx \longrightarrow \frac{\mathrm{Log}[c+d\,x]}{2\,a\,d}+\frac{1}{2\,a}\int \frac{\mathrm{Cos}[2\,e+2\,f\,x]}{c+d\,x}\,dx+\frac{1}{2\,b}\int \frac{\mathrm{Sin}[2\,e+2\,f\,x]}{c+d\,x}\,dx$$

Program code:

```
Int[1/((c_.+d_.*x_)*(a_+b_.*tan[e_.+f_.*x_])),x_Symbol] :=
  Log[c+d*x]/(2*a*d) +
  1/(2*a)*Int[Cos[2*e+2*f*x]/(c+d*x),x] +
  1/(2*b)*Int[Sin[2*e+2*f*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2+b^2,0]
```

4: $\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2==0 \land m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $a^2+b^2==0$, then $\frac{1}{a+b\,\mathrm{Tan}[z]}==\frac{1}{2\,a}+\frac{e^{\frac{2\,a\,z}{b}}}{2\,a}$

Rule: If $a^2+b^2==0 \land m\notin\mathbb{Z}$, then

$$\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx \to \frac{(c+d\,x)^{m+1}}{2\,a\,d\,(m+1)}+\frac{1}{2\,a}\int (c+d\,x)^m\,e^{\frac{2\,a}{b}(e+f\,x)}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (c+d*x)^(m+1)/(2*a*d*(m+1)) +
  1/(2*a)*Int[(c+d*x)^m*E^(2*a/b*(e+f*x)),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[a^2+b^2,0] && Not[IntegerQ[m]]
```

2: $\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\land(m\mid n)\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $a^2+b^2==0$, then $\frac{1}{a+b\,\mathrm{Tan}[z]}==\frac{1}{2\,a}+\frac{\mathrm{Cos}[2\,z]}{2\,a}+\frac{\mathrm{Sin}[2\,z]}{2\,b}$

Rule: If $a^2+b^2==0\land(m\mid n)\in\mathbb{Z}^-$, then

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow\,\int (c+d\,x)^m\,\mathrm{ExpandIntegrand}\!\left[\left(\frac{1}{2\,a}+\frac{\mathrm{Cos}[2\,e+2\,f\,x]}{2\,a}+\frac{\mathrm{Sin}[2\,e+2\,f\,x]}{2\,b}\right)^{-n},\,x\right]dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,(1/(2*a)+Cos[2*e+2*f*x]/(2*a)+Sin[2*e+2*f*x]/(2*b))^(-n),x],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2+b^2,0] && ILtQ[m,0] && ILtQ[n,0]
```

3: $\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2==0\,\land\,n\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $a^2+b^2==0$, then $\frac{1}{a+b\,\mathrm{Tan}[z]}==\frac{1}{2\,a}+\frac{e^{\frac{2\,a\,z}{b}}}{2\,a}$

Rule: If $a^2+b^2==0\,\land\,n\in\mathbb{Z}^-$, then

$$\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow\,\int (c+d\,x)^m\,\mathrm{ExpandIntegrand}\left[\left(\frac{1}{2\,a}+\frac{e^{\frac{2\,a}{b}\,(e+f\,x)}}{2\,a}\right)^{-n},\,x\right]dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,(1/(2*a)+E^(2*a/b*(e+f*x))/(2*a))^(-n),x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[a^2+b^2,0] && ILtQ[n,0]
```

4: $\int (c+d x)^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $a^2+b^2==0 \land n+1 \in \mathbb{Z}^- \land m>0$

Derivation: Integration by parts

Note: If $a^2+b^2==0 \land n \in \mathbb{Z}^-$, then $\int (a+b \operatorname{Tan}[e+f x])^n \, dx$ is a monomial in x plus terms of the form $g (a+b \operatorname{Tan}[e+f x])^k$ where $n \leq k<0$.

Rule: If $a^2+b^2==0 \land n+1 \in \mathbb{Z}^- \land m>0$, let $u=\int (a+b \operatorname{Tan}[e+f x])^n \, dx$, then

$$\int (c+d x)^m (a+b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow u (c+d x)^m-d m \int u (c+d x)^{m-1} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  With[{u=IntHide[(a+b*Tan[e+f*x])^n,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[Dist[(c+d*x)^(m-1),u,x],x]] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2+b^2,0] && ILtQ[n,-1] && GtQ[m,0]
```

4. $\int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2\neq 0\wedge n\in\mathbb{Z}^-\wedge m\in\mathbb{Z}^+$

1: $\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2\neq 0\wedge m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\,\mathrm{Tan}[z]}==\frac{1}{a+i\,b}+\frac{2\,i\,b\,e^{2\,i\,z}}{(a+i\,b)^2+(a^2+b^2)\,e^{2\,i\,z}}$

Rule: If $a^2+b^2\neq 0\wedge m\in\mathbb{Z}^+$, then

$$\int \frac{(c+d\,x)^m}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx\to\frac{(c+d\,x)^{m+1}}{d\,(m+1)\,(a+i\,b)}+2\,i\,b\int \frac{(c+d\,x)^m\,e^{2\,i\,(e+f\,x)}}{(a+i\,b)^2+(a^2+b^2)\,e^{2\,i\,(e+f\,x)}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*tan[e_.+k_.*Pi+f_.*x_]),x_Symbol] :=
  (c+d*x)^(m+1)/(d*(m+1)*(a+I*b)) +
  2*I*b*Int[(c+d*x)^m*E^(2*I*k*Pi)*E^Simp[2*I*(e+f*x),x]/((a+I*b)^2+(a^2+b^2)*E^(2*I*k*Pi)*E^Simp[2*I*(e+f*x),x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IntegerQ[4*k] && NeQ[a^2+b^2,0] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (c+d*x)^(m+1)/(d*(m+1)*(a+I*b)) +
  2*I*b*Int[(c+d*x)^m*E^Simp[2*I*(e+f*x),x]/((a+I*b)^2+(a^2+b^2)*E^Simp[2*I*(e+f*x),x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a^2+b^2,0] && IGtQ[m,0]
```

2: $\int \frac{(c+d\,x)}{(a+b\,\mathrm{Tan}[e+f\,x])^2}\,dx$ when $a^2+b^2\neq 0$

Rule: If $a^2+b^2\neq 0$, then

$$\int \frac{c+d\,x}{(a+b\,\text{Tan}[e+f\,x])^2}\,dx \longrightarrow -\frac{(c+d\,x)^2}{2\,d\,(a^2+b^2)}-\frac{b\,(c+d\,x)}{f\,(a^2+b^2)\,(a+b\,\text{Tan}[e+f\,x])}+\frac{1}{f\,(a^2+b^2)}\int \frac{b\,d+2\,a\,c\,f+2\,a\,d\,f\,x}{a+b\,\text{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)/(a_+b_.*tan[e_.+f_.*x_])^2,x_Symbol] :=
  -(c+d*x)^2/(2*d*(a^2+b^2)) -
  b*(c+d*x)/(f*(a^2+b^2)*(a+b*Tan[e+f*x])) +
  1/(f*(a^2+b^2))*Int[(b*d+2*a*c*f+2*a*d*f*x)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a^2+b^2,0]
```

3: $\int (c+d\,x)^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$ when $a^2+b^2\neq 0\,\land\,n\in\mathbb{Z}^-\,\land\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\,\text{Tan}[z]} == \frac{1}{a-i\,b}-\frac{2\,i\,b}{a^2+b^2+(a-i\,b)^2\,e^{2\,i\,z}}$

Basis: $\frac{1}{a+b\,\text{Cot}[z]} == \frac{1}{a+i\,b}+\frac{2\,i\,b}{a^2+b^2-(a+i\,b)^2\,e^{2\,i\,z}}$

Rule: If $a^2+b^2\neq 0\,\land\,n\in\mathbb{Z}^-\,\land\,m\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx \longrightarrow \int (c+d\,x)^m\,\text{ExpandIntegrand}\Big[\left(\frac{1}{a-i\,b}-\frac{2\,i\,b}{a^2+b^2+(a-i\,b)^2\,e^{2\,i\,(e+f\,x)}}\right)^{-n},\,x\Big]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,(1/(a-I*b)-2*I*b/(a^2+b^2+(a-I*b)^2*E^(2*I*(e+f*x))))^(-n),x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a^2+b^2,0] && ILtQ[n,0] && IGtQ[m,0]
```

X: $\int (c+d\,x)^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$

Basis: $\text{Tan}[e+f\,x] = -\text{Cot}\left[e-\frac{\pi}{2}+f\,x\right]$

Basis: $\text{Tan}[e+f\,x] = i\,\text{Tanh}[-i\,e-i\,f\,x]$

Basis: $\text{Tan}[e+f\,x] = i\,\text{Coth}\left[-i\left(e-\frac{\pi}{2}\right)-i\,f\,x\right]$

Rule:

$$\int (c+d\,x)^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx \to \int (c+d\,x)^m\,(a+b\,\text{Tan}[e+f\,x])^n\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*tan[e_.+f_.*x_]^n_.,x_Symbol] :=
  If[MatchQ[f,f1_.*Complex[0,j_]],
    If[MatchQ[e,e1_.+Pi/2],
      I^n*Unintegrable[(c+d*x)^m*Coth[-I*(e-Pi/2)-I*f*x]^n,x],
    I^n*Unintegrable[(c+d*x)^m*Tanh[-I*e-I*f*x]^n,x]],
  If[MatchQ[e,e1_.+Pi/2],
    (-1)^n*Unintegrable[(c+d*x)^m*Cot[e-Pi/2+f*x]^n,x],
  Unintegrable[(c+d*x)^m*Tan[e+f*x]^n,x]]] /;
FreeQ[{c,d,e,f,m,n},x] && IntegerQ[n]
```

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*(a+b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

N: $\int u^m\,(a+b\,\mathrm{Tan}[v])^n\,dx$ when $u == c+d\,x \land v == e+f\,x$

Derivation: Algebraic normalization

Rule: If $u == c+d\,x \land v == e+f\,x$, then

$$\int u^m\,(a+b\,\mathrm{Tan}[v])^n\,dx \longrightarrow \int (c+d\,x)^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$$

Program code:

```
Int[u_^m_.*(a_.+b_.*Tan[v_])^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*(a+b*Tan[ExpandToSum[v,x]])^n,x] /;
FreeQ[{a,b,m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

```
Int[u_^m_.*(a_.+b_.*Cot[v_])^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*(a+b*Cot[ExpandToSum[v,x]])^n,x] /;
FreeQ[{a,b,m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```