Rules for integrands of the form $(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])$

1: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c+a\,d==0\,\land\,a^2+b^2==0$

Derivation: Integration by substitution

Basis: If $b\,c+a\,d==0\,\land\,a^2+b^2==0$, then
$(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n==$
$\frac{a\,c}{f}\,\mathrm{Subst}\left[(a+b\,x)^{m-1}\,(c+d\,x)^{n-1},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\mathrm{Tan}[e+f\,x]$

Rule: If $b\,c+a\,d==0\,\land\,a^2+b^2==0$, then

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\,\rightarrow\,\frac{a\,c}{f}\,\mathrm{Subst}\left[\int(a+b\,x)^{m-1}\,(c+d\,x)^{n-1}\,(A+B\,x)\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_+d_.*tan[e_.+f_.*x_])^n_.*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  a*c/f*Subst[Int[(a+b*x)^(m-1)*(c+d*x)^(n-1)*(A+B*x),x],x,Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2+b^2,0]
```

2. $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq0$

1. $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq0\,\land\,m\leq-1$

1: $\int\frac{(c+d\,\mathrm{Tan}[e+f\,x])\,(A+B\,\mathrm{Tan}[e+f\,x])}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq0$

Derivation: Algebraic expansion

Basis: $\frac{(c+d\,z)\,(A+B\,z)}{a+b\,z}==\frac{B\,d\,z}{b}+\frac{A\,b\,c+(A\,b\,d+B\,(b\,c-a\,d))\,z}{b\,(a+b\,z)}$

Rule: If $b\,c-a\,d\neq0$, then

$$\int \frac{(c + d \operatorname{Tan}[e + f x]) (A + B \operatorname{Tan}[e + f x])}{a + b \operatorname{Tan}[e + f x]} dx \rightarrow \frac{B d}{b} \int \operatorname{Tan}[e + f x] dx + \frac{1}{b} \int \frac{A b c + (A b d + B (b c - a d)) \operatorname{Tan}[e + f x]}{a + b \operatorname{Tan}[e + f x]} dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])*(A_.+B_.*tan[e_.+f_.*x_])/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  B*d/b*Int[Tan[e+f*x],x] + 1/b*Int[Simp[A*b*c+(A*b*d+B*(b*c-a*d))*Tan[e+f*x],x]/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0]
```

2. $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) (A + B \operatorname{Tan}[e + f x]) dx$ when $b c - a d \neq 0 \land m < -1$

 1: $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) (A + B \operatorname{Tan}[e + f x]) dx$ when $b c - a d \neq 0 \land m < -1 \land a^2 + b^2 == 0$

Derivation: Symmetric tangent recurrence 2a with n → 1 and ???

Rule: If $b c - a d \neq 0 \land m < -1 \land a^2 + b^2 == 0$, then

$$\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) (A + B \operatorname{Tan}[e + f x]) dx \longrightarrow$$

$$-\frac{(A b - a B) (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])}{2 a f m} +$$

$$\frac{1}{2 a^2 m} \int (a + b \operatorname{Tan}[e + f x])^{m+1} (A (b d + a c m) - B (a d + b c m) - d (b B (m - 1) - a A (m + 1)) \operatorname{Tan}[e + f x]) dx \longrightarrow$$

$$-\frac{(A b - a B) (a c + b d) (a + b \operatorname{Tan}[e + f x])^m}{2 a^2 f m} + \frac{1}{2 a b} \int (a + b \operatorname{Tan}[e + f x])^{m+1} (A b c + a B c + a A d + b B d + 2 a B d \operatorname{Tan}[e + f x]) dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*(a*c+b*d)*(a+b*Tan[e+f*x])^m/(2*a^2*f*m) +
  1/(2*a*b)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[A*b*c+a*B*c+a*A*d+b*B*d+2*a*B*d*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] && EqQ[a^2+b^2,0]
```

2: $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) (A + B \operatorname{Tan}[e + f x]) \, dx$ when $b c - a d \neq 0 \land m < -1 \land a^2 + b^2 \neq 0$

Derivation: Tangent recurrence 1b with $A \to A c$, $B \to B c + A d$, $C \to B d$, $n \to 0$

Rule: If $b c - a d \neq 0 \land m < -1 \land a^2 + b^2 \neq 0$, then

$$\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) (A + B \operatorname{Tan}[e + f x]) \, dx \longrightarrow$$

$$\frac{(b c - a d) (A b - a B) (a + b \operatorname{Tan}[e + f x])^{m+1}}{b f (m + 1) (a^2 + b^2)} + \frac{1}{a^2 + b^2} \int (a + b \operatorname{Tan}[e + f x])^{m+1} (a A c + b B c + A b d - a B d - (A b c - a B c - a A d - b B d) \operatorname{Tan}[e + f x]) \, dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (b*c-a*d)*(A*b-a*B)*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)*(a^2+b^2)) +
  1/(a^2+b^2)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[a*A*c+b*B*c+A*b*d-a*B*d-(A*b*c-a*B*c-a*A*d-b*B*d)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] && NeQ[a^2+b^2,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\nleq -1$

Derivation: Tangent recurrence 2b with $A\to A\,c,\ B\to B\,c+A\,d,\ C\to B\,d,\ n\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m\nleq -1$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{B\,d\,(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}}{b\,f\,(m+1)}+\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(A\,c-B\,d+(B\,c+A\,d)\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  B*d*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)) +
  Int[(a+b*Tan[e+f*x])^m*Simp[A*c-B*d+(B*c+A*d)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && Not[LeQ[m,-1]]
```

3. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$

1. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m>1\,\wedge\,n<-1$

Derivation: Symmetric tangent recurrence 1a

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m>1\,\wedge\,n<-1$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{a^2\,(B\,c-A\,d)\,(a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d\,f\,(b\,c+a\,d)\,(n+1)}-\frac{a}{d\,(b\,c+a\,d)\,(n+1)}$$

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}\,\left(A\,b\,d\,(m-n-2)-B\,(b\,c\,(m-1)+a\,d\,(n+1))+(a\,A\,d\,(m+n)-B\,(a\,c\,(m-1)+b\,d\,(n+1)))\,\mathrm{Tan}[e+f\,x]\right)\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -a^2*(B*c-A*d)*(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(b*c+a*d)*(n+1)) -
  a/(d*(b*c+a*d)*(n+1))*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)*
    Simp[A*b*d*(m-n-2)-B*(b*c*(m-1)+a*d*(n+1))+(a*A*d*(m+n)-B*(a*c*(m-1)+b*d*(n+1)))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && GtQ[m,1] && LtQ[n,-1]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,m>1\,\land\,n\not<-1$

Derivation: Symmetric tangent recurrence 1b

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,m>1\,\land\,n\not<-1$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\;\longrightarrow$$

$$\frac{b\,B\,(a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d\,f\,(m+n)}+$$

$$\frac{1}{d\,(m+n)}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n}\,\left(a\,A\,d\,(m+n)+B\,(a\,c\,(m-1)-b\,d\,(n+1))-(B\,(b\,c-a\,d)\,(m-1)-d\,(A\,b+a\,B)\,(m+n))\,\mathrm{Tan}[e+f\,x]\right)\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  b*B*(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n)) +
  1/(d*(m+n))*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^n*
    Simp[a*A*d*(m+n)+B*(a*c*(m-1)-b*d*(n+1))-(B*(b*c-a*d)*(m-1)-d*(A*b+a*B)*(m+n))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && GtQ[m,1] && Not[LtQ[n,-1]]
```

2. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,m<0$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,m<0\,\land\,n>0$

Derivation: Symmetric tangent recurrence 2a

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,m<0\,\land\,n>0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\longrightarrow$$

$$-\frac{(A\,b-a\,B)\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n}{2\,a\,f\,m}+$$

$$\frac{1}{2\,a^2\,m}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n-1}\,(A\,(a\,c\,m+b\,d\,n)-B\,(b\,c\,m+a\,d\,n)-d\,(b\,B\,(m-n)-a\,A\,(m+n))\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n/(2*a*f*m) +
  1/(2*a^2*m)*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n-1)*
    Simp[A*(a*c*m+b*d*n)-B*(b*c*m+a*d*n)-d*(b*B*(m-n)-a*A*(m+n))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && LtQ[m,0] && GtQ[n,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,m<0\,\land\,n\not>0$

Derivation: Symmetric tangent recurrence 2b

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,m<0\,\land\,n\not>0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\longrightarrow$$

$$\frac{(a\,A+b\,B)\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{2\,f\,m\,(b\,c-a\,d)}+$$

$$\frac{1}{2\,a\,m\,(b\,c-a\,d)}\int(a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^{n}\,(A\,(b\,c\,m-a\,d\,(2\,m+n+1))+B\,(a\,c\,m-b\,d\,(n+1))+d\,(A\,b-a\,B)\,(m+n+1)\,\text{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (a*A+b*B)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(2*f*m*(b*c-a*d)) +
  1/(2*a*m*(b*c-a*d))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
    Simp[A*(b*c*m-a*d*(2*m+n+1))+B*(a*c*m-b*d*(n+1))+d*(A*b-a*B)*(m+n+1)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && LtQ[m,0] && Not[GtQ[n,0]]
```

3: $\int(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,n>0$

Derivation: Symmetric tangent recurrence 3a

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,n>0$, then

$$\int(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{B\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n}{f\,(m+n)}+$$

$$\frac{1}{a\,(m+n)}\int(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^{n-1}\,(a\,A\,c\,(m+n)-B\,(b\,c\,m+a\,d\,n)+(a\,A\,d\,(m+n)-B\,(b\,d\,m-a\,c\,n))\,\text{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  B*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n/(f*(m+n)) +
  1/(a*(m+n))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n-1)*
    Simp[a*A*c*(m+n)-B*(b*c*m+a*d*n)+(a*A*d*(m+n)-B*(b*d*m-a*c*n))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && GtQ[n,0]
```

4: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,n<-1$

Derivation: Symmetric tangent recurrence 3b

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,n<-1$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(A\,d-B\,c)\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{f\,(n+1)\,(c^2+d^2)}-$$

$$\frac{1}{a\,(n+1)\,(c^2+d^2)}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}\,(A\,(b\,d\,m-a\,c\,(n+1))-B\,(b\,c\,m+a\,d\,(n+1))-a\,(B\,c-A\,d)\,(m+n+1)\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (A*d-B*c)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(f*(n+1)*(c^2+d^2)) -
  1/(a*(n+1)*(c^2+d^2))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)*
    Simp[A*(b*d*m-a*c*(n+1))-B*(b*c*m+a*d*(n+1))-a*(B*c-A*d)*(m+n+1)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && LtQ[n,-1]
```

5: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,A\,b+a\,B==0$

Derivation: Integration by substitution

Basis: If $a^2+b^2==0\,\land\,A\,b+a\,B==0$, then
$(a+b\,\mathrm{Tan}[e+f\,x])^m\,(A+B\,\mathrm{Tan}[e+f\,x])==\frac{b\,B}{f}\,\mathrm{Subst}\left[(a+b\,x)^{m-1},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\mathrm{Tan}[e+f\,x]$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,A\,b+a\,B==0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\;\rightarrow\;\frac{b\,B}{f}\,\mathrm{Subst}\left[\int (a+b\,x)^{m-1}\,(c+d\,x)^n\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  b*B/f*Subst[Int[(a+b*x)^(m-1)*(c+d*x)^n,x],x,Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && EqQ[A*b+a*B,0]
```

6. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,A\,b+a\,B\neq 0$

1: $\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^m\,(A+B\,\mathrm{Tan}[e+f\,x])}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,A\,b+a\,B\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z}{c+d\,z}==\frac{A\,b+a\,B}{b\,c+a\,d}-\frac{(B\,c-A\,d)\,(a-b\,z)}{(b\,c+a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,A\,b+a\,B\neq 0$, then

$$\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^m\,(A+B\,\mathrm{Tan}[e+f\,x])}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx\,\rightarrow\,\frac{A\,b+a\,B}{b\,c+a\,d}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,dx-\frac{B\,c-A\,d}{b\,c+a\,d}\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^m\,(a-b\,\mathrm{Tan}[e+f\,x])}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(A_.+B_.*tan[e_.+f_.*x_])/(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (A*b+a*B)/(b*c+a*d)*Int[(a+b*Tan[e+f*x])^m,x] -
  (B*c-A*d)/(b*c+a*d)*Int[(a+b*Tan[e+f*x])^m*(a-b*Tan[e+f*x])/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[A*b+a*B,0]
```

x: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\land a^2+b^2==0$

Derivation: Algebraic expansion

Baisi: $A+B\,z==\frac{A\,b-a\,B}{b}+\frac{B\,(a+b\,z)}{b}$

Rule: If $b\,c-a\,d\neq 0\land a^2+b^2==0\land c^2+d^2\neq 0$, then

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{A\,b-a\,B}{b}\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx+\frac{B}{b}\int (a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$$

Program code:

```
(* Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)/b*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n,x] +
  B/b*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,A,B,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] *)
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,A\,b+a\,B\neq 0$

Derivation: Algebraic expansion

Basis: $A+B\,z==\frac{A\,b+a\,B}{b}-\frac{B\,(a-b\,z)}{b}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,A\,b+a\,B\neq 0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{A\,b+a\,B}{b}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx-\frac{B}{b}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(a-b\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (A*b+a*B)/b*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n,x] -
  B/b*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n*(a-b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[A*b+a*B,0]
```

4. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$

1. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,\neg\,(2\,m\mid 2\,n)\in\mathbb{Z}$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,\neg\,(2\,m\mid 2\,n)\in\mathbb{Z}\,\land\,A^2+B^2==0$

Derivation: Integration by substitution

Basis: If $A^2+B^2==0$, then $A+B\,\mathrm{Tan}[e+f\,x]==\frac{A^2}{f}\,\mathrm{Subst}\left[\frac{1}{A-B\,x},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\mathrm{Tan}[e+f\,x]$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,\neg\,(2\,m\mid 2\,n)\in\mathbb{Z}\,\land\,A^2+B^2==0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx \to \frac{A^2}{f}\,\mathrm{Subst}\Big[\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{A-B\,x}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\Big]$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  A^2/f*Subst[Int[(a+b*x)^m*(c+d*x)^n/(A-B*x),x],x,Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] &&
  Not[IntegersQ[2*m,2*n]] && EqQ[A^2+B^2,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,\neg\,(2\,m\mid 2\,n)\in\mathbb{Z}\,\land\,A^2+B^2\neq 0$

Derivation: Algebraic expansion

Basis: $A+B\,z == \frac{A+i\,B}{2}\,(1-i\,z)+\frac{A-i\,B}{2}\,(1+i\,z)$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,\neg\,(2\,m\mid 2\,n)\in\mathbb{Z}\,\land\,A^2+B^2\neq 0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx \to$$
$$\frac{A+i\,B}{2}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(1-i\,\mathrm{Tan}[e+f\,x])\,dx+\frac{A-i\,B}{2}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(1+i\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (A+I*B)/2*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n*(1-I*Tan[e+f*x]),x] +
  (A-I*B)/2*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n*(1+I*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] &&
  Not[IntegersQ[2*m,2*n]] && NeQ[A^2+B^2,0]
```

2. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m>1$

   1: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m>1\,\land\,n<-1$

Derivation: Tangent recurrence 1a with $A\to a\,A,\ B\ {-}{>}\ A\,b+a\,B,\ C\to b\,B,\ m\to m-1$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m>1\,\land\,n<-1$, then

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx\ \longrightarrow$$

$$\frac{(b\,c-a\,d)\,(B\,c-A\,d)\,(a+b\,\text{Tan}[e+f\,x])^{m-1}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,(c^2+d^2)}-$$

$$\frac{1}{d\,(n+1)\,(c^2+d^2)}\int (a+b\,\text{Tan}[e+f\,x])^{m-2}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}\cdot$$

$$\big(a\,A\,d\,(b\,d\,(m-1)-a\,c\,(n+1))+(b\,B\,c-(A\,b+a\,B)\,d)\,(b\,c\,(m-1)+a\,d\,(n+1))-$$
$$d\,((a\,A-b\,B)\,(b\,c-a\,d)+(A\,b+a\,B)\,(a\,c+b\,d))\,(n+1)\,\text{Tan}[e+f\,x]-b\,(d\,(A\,b\,c+a\,B\,c-a\,A\,d)\,(m+n)-b\,B\,(c^2\,(m-1)-d^2\,(n+1)))\,\text{Tan}[e+f\,x]^2\big)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (b*c-a*d)*(B*c-A*d)*(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+1)*(c^2+d^2)) -
  1/(d*(n+1)*(c^2+d^2))*Int[(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^(n+1)*
    Simp[a*A*d*(b*d*(m-1)-a*c*(n+1))+(b*B*c-(A*b+a*B)*d)*(b*c*(m-1)+a*d*(n+1))-
      d*((a*A-b*B)*(b*c-a*d)+(A*b+a*B)*(a*c+b*d))*(n+1)*Tan[e+f*x]-
      b*(d*(A*b*c+a*B*c-a*A*d)*(m+n)-b*B*(c^2*(m-1)-d^2*(n+1)))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,1] && LtQ[n,-1] &&
  (IntegerQ[m] || IntegersQ[2*m,2*n])
```

2: $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land m > 1 \land n \not< -1$

Derivation: Tangent recurrence 2a with $A \to a A$, $B \to A b + a B$, $C \to b B$, $m \to m - 1$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land m > 1 \land n \not< -1$, then

$$\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x]) \, dx \longrightarrow$$

$$\frac{b B (a + b \operatorname{Tan}[e + f x])^{m-1} (c + d \operatorname{Tan}[e + f x])^{n+1}}{d f (m + n)} + \frac{1}{d (m + n)} \int (a + b \operatorname{Tan}[e + f x])^{m-2} (c + d \operatorname{Tan}[e + f x])^n \cdot$$
$$\left(a^2 A d (m + n) - b B (b c (m - 1) + a d (n + 1)) + d (m + n) (2 a A b + B (a^2 - b^2)) \operatorname{Tan}[e + f x] - (b B (b c - a d) (m - 1) - b (A b + a B) d (m + n)) \operatorname{Tan}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  b*B*(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n)) +
  1/(d*(m+n))*Int[(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^n*
    Simp[a^2*A*d*(m+n)-b*B*(b*c*(m-1)+a*d*(n+1))+
      d*(m+n)*(2*a*A*b+B*(a^2-b^2))*Tan[e+f*x]-
      (b*B*(b*c-a*d)*(m-1)-b*(A*b+a*B)*d*(m+n))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,1] &&
  (IntegerQ[m] || IntegersQ[2*m,2*n]) && Not[IGtQ[n,1] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

3. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1$

1: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$

Derivation: Tangent recurrence 1b with $C\to 0$

Derivation: Tangent recurrence 3a with $A\to A\,c,\ B\to B\,c+A\,d,\ C\to B\,d,\ n\to n-1$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<1$, then

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{(A\,b-a\,B)\,(a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^n}{f\,(m+1)\,(a^2+b^2)}+$$
$$\frac{1}{b\,(m+1)\,(a^2+b^2)}\int (a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^{n-1}\cdot$$
$$(b\,B\,(b\,c\,(m+1)+a\,d\,n)+A\,b\,(a\,c\,(m+1)-b\,d\,n)-b\,(A\,(b\,c-a\,d)-B\,(a\,c+b\,d))\,(m+1)\,\text{Tan}[e+f\,x]-b\,d\,(A\,b-a\,B)\,(m+n+1)\,\text{Tan}[e+f\,x]^2)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)*(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n/(f*(m+1)*(a^2+b^2)) +
  1/(b*(m+1)*(a^2+b^2))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n-1)*
    Simp[b*B*(b*c*(m+1)+a*d*n)+A*b*(a*c*(m+1)-b*d*n)-b*(A*(b*c-a*d)-B*(a*c+b*d))*(m+1)*Tan[e+f*x]-b*d*(A*b-a*B)*(m+n+1)*Tan[e+f*x]^2,x
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,-1] && LtQ[0,n,1] && (IntegerQ[m] || Integer
```

2: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,n\not>0$

Derivation: Tangent recurrence 3a with $C\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1$, then

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx\;\longrightarrow$$
$$\frac{b\,(A\,b-a\,B)\,(a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}}{f\,(m+1)\,(b\,c-a\,d)\,(a^2+b^2)}+$$
$$\frac{1}{(m+1)\,(b\,c-a\,d)\,(a^2+b^2)}\int (a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^n\cdot$$
$$\left(b\,B\,(b\,c\,(m+1)+a\,d\,(n+1))+A\,\left(a\,(b\,c-a\,d)\,(m+1)-b^2\,d\,(m+n+2)\right)-(A\,b-a\,B)\,(b\,c-a\,d)\,(m+1)\,\text{Tan}[e+f\,x]-b\,d\,(A\,b-a\,B)\,(m+n+2)\,\text{Tan}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  b*(A*b-a*B)*(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n+1)/(f*(m+1)*(b*c-a*d)*(a^2+b^2)) +
  1/((m+1)*(b*c-a*d)*(a^2+b^2))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
    Simp[b*B*(b*c*(m+1)+a*d*(n+1))+A*(a*(b*c-a*d)*(m+1)-b^2*d*(m+n+2)) -
      (A*b-a*B)*(b*c-a*d)*(m+1)*Tan[e+f*x] -
      b*d*(A*b-a*B)*(m+n+2)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,-1] && (IntegerQ[m] || IntegersQ[2*m,2*n])
  Not[ILtQ[n,-1] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

4: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,0<m<1\,\land\,0<n<1$

Derivation: Tangent recurrence 2a with $A\to A\,c,\ B\to B\,c+A\,d,\ C\to B\,d,\ n\to n-1$

Derivation: Tangent recurrence 2b with $A\to a\,A,\ B\to A\,b+a\,B,\ C\to b\,B,\ m\to m-1$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,0<m<1\,\land\,0<n<1$, then

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{B\,(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n}{f\,(m+n)}+$$
$$\frac{1}{m+n}\int (a+b\,\text{Tan}[e+f\,x])^{m-1}\,(c+d\,\text{Tan}[e+f\,x])^{n-1}\cdot$$
$$\left(a\,A\,c\,(m+n)-B\,(b\,c\,m+a\,d\,n)+(A\,b\,c+a\,B\,c+a\,A\,d-b\,B\,d)\,(m+n)\,\text{Tan}[e+f\,x]+(A\,b\,d\,(m+n)+B\,(a\,d\,m+b\,c\,n))\,\text{Tan}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  B*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n/(f*(m+n)) +
  1/(m+n)*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n-1)*
    Simp[a*A*c*(m+n)-B*(b*c*m+a*d*n)+(A*b*c+a*B*c+a*A*d-b*B*d)*(m+n)*Tan[e+f*x]+(A*b*d*(m+n)+B*(a*d*m+b*c*n))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[0,m,1] && LtQ[0,n,1]
```

5. $\int \frac{(c+d\,\text{Tan}[e+f\,x])^n\,(A+B\,\text{Tan}[e+f\,x])}{a+b\,\text{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$

1: $\int \frac{A+B\,\text{Tan}[e+f\,x]}{(a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z}{(a+b z)(c+d z)} == \frac{B(b c+a d)+A(a c-b d)}{\left(a^2+b^2\right)\left(c^2+d^2\right)} + \frac{b(A b-a B)(b-a z)}{\left(a^2+b^2\right)(b c-a d)(a+b z)} + \frac{d(B c-A d)(d-c z)}{(b c-a d)\left(c^2+d^2\right)(c+d z)}$

Rule: If $b c-a d \neq 0 \wedge a^2+b^2 \neq 0 \wedge c^2+d^2 \neq 0$, then

$$\int \frac{A+B \operatorname{Tan}[e+f x]}{(a+b \operatorname{Tan}[e+f x])(c+d \operatorname{Tan}[e+f x])} \, dx \longrightarrow$$

$$\frac{(B(b c+a d)+A(a c-b d)) x}{\left(a^2+b^2\right)\left(c^2+d^2\right)}+\frac{b(A b-a B)}{(b c-a d)\left(a^2+b^2\right)} \int \frac{b-a \operatorname{Tan}[e+f x]}{a+b \operatorname{Tan}[e+f x]} \, dx+\frac{d(B c-A d)}{(b c-a d)\left(c^2+d^2\right)} \int \frac{d-c \operatorname{Tan}[e+f x]}{c+d \operatorname{Tan}[e+f x]} \, dx$$

Program code:

```
Int[(A_.+B_.*tan[e_.+f_.*x_])/((a_+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])),x_Symbol] :=
  (B*(b*c+a*d)+A*(a*c-b*d))*x/((a^2+b^2)*(c^2+d^2)) +
  b*(A*b-a*B)/((b*c-a*d)*(a^2+b^2))*Int[(b-a*Tan[e+f*x])/(a+b*Tan[e+f*x]),x] +
  d*(B*c-A*d)/((b*c-a*d)*(c^2+d^2))*Int[(d-c*Tan[e+f*x])/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

2: $\int \frac{\sqrt{c + d \operatorname{Tan}[e + f x]}\, (A + B \operatorname{Tan}[e + f x])}{a + b \operatorname{Tan}[e + f x]}\, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{c+d z}\, (A+B z)}{a+b z} == \frac{A (a c+b d)+B (b c-a d)-(A (b c-a d)-B (a c+b d)) z}{(a^2+b^2) \sqrt{c+d z}} - \frac{(b c-a d) (B a-A b) (1+z^2)}{(a^2+b^2) (a+b z) \sqrt{c+d z}}$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0$, then

$$\int \frac{\sqrt{c + d \operatorname{Tan}[e + f x]}\, (A + B \operatorname{Tan}[e + f x])}{a + b \operatorname{Tan}[e + f x]}\, dx \to$$

$$\frac{1}{a^2 + b^2} \int \frac{A (a c + b d) + B (b c - a d) - (A (b c - a d) - B (a c + b d)) \operatorname{Tan}[e + f x]}{\sqrt{c + d \operatorname{Tan}[e + f x]}}\, dx - \frac{(b c - a d) (B a - A b)}{a^2 + b^2} \int \frac{1 + \operatorname{Tan}[e + f x]^2}{(a + b \operatorname{Tan}[e + f x]) \sqrt{c + d \operatorname{Tan}[e + f x]}}\, dx$$

Program code:

```
Int[Sqrt[c_.+d_.*tan[e_.+f_.*x_]]*(A_.+B_.*tan[e_.+f_.*x_])/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(a^2+b^2)*Int[Simp[A*(a*c+b*d)+B*(b*c-a*d)-(A*(b*c-a*d)-B*(a*c+b*d))*Tan[e+f*x],x]/Sqrt[c+d*Tan[e+f*x]],x] -
  (b*c-a*d)*(B*a-A*b)/(a^2+b^2)*Int[(1+Tan[e+f*x]^2)/((a+b*Tan[e+f*x])*Sqrt[c+d*Tan[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

3: $\int \frac{(c + d \operatorname{Tan}[e + f x])^n\, (A + B \operatorname{Tan}[e + f x])}{a + b \operatorname{Tan}[e + f x]}\, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z}{a+b z} == \frac{a A+b B-(A b-a B) z}{a^2+b^2} + \frac{b (A b-a B) (1+z^2)}{(a^2+b^2) (a+b z)}$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0$, then

$$\int \frac{(c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x])}{a + b \operatorname{Tan}[e + f x]} \, dx \longrightarrow$$

$$\frac{1}{a^2 + b^2} \int (c + d \operatorname{Tan}[e + f x])^n (a A + b B - (A b - a B) \operatorname{Tan}[e + f x]) \, dx + \frac{b (A b - a B)}{a^2 + b^2} \int \frac{(c + d \operatorname{Tan}[e + f x])^n \left(1 + \operatorname{Tan}[e + f x]^2\right)}{a + b \operatorname{Tan}[e + f x]} \, dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_])/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(a^2+b^2)*Int[(c+d*Tan[e+f*x])^n*Simp[a*A+b*B-(A*b-a*B)*Tan[e+f*x],x],x] +
  b*(A*b-a*B)/(a^2+b^2)*Int[(c+d*Tan[e+f*x])^n*(1+Tan[e+f*x]^2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

6: $\int \frac{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,(A+B\,\mathrm{Tan}[e+f\,x])}{\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\ \land\ a^2+b^2\neq 0\ \land\ c^2+d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\sqrt{a+b\,z}\,(A+B\,z) == \frac{a\,A-b\,B+(A\,b+a\,B)\,z}{\sqrt{a+b\,z}}+\frac{b\,B\,(1+z^2)}{\sqrt{a+b\,z}}$

Note: This rule should be generalized for all integrands of the form $\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])$ when $A\,b-a\,B\neq 0\ \land\ a^2+b^2\neq 0$.

Rule: If $b\,c-a\,d\neq 0\ \land\ a^2+b^2\neq 0\ \land\ c^2+d^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,(A+B\,\mathrm{Tan}[e+f\,x])}{\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx \to \int \frac{a\,A-b\,B+(A\,b+a\,B)\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx+b\,B\int \frac{1+\mathrm{Tan}[e+f\,x]^2}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[a_.+b_.*tan[e_.+f_.*x_]]*(A_.+B_.*tan[e_.+f_.*x_])/Sqrt[c_.+d_.*tan[e_.+f_.*x_]],x_Symbol] :=
  Int[Simp[a*A-b*B+(A*b+a*B)*Tan[e+f*x],x]/(Sqrt[a+b*Tan[e+f*x]]*Sqrt[c+d*Tan[e+f*x]]),x] +
  b*B*Int[(1+Tan[e+f*x]^2)/(Sqrt[a+b*Tan[e+f*x]]*Sqrt[c+d*Tan[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

x. $\int \frac{A + B \operatorname{Tan}[e + f x]}{\sqrt{a + b \operatorname{Tan}[e + f x]} \sqrt{c + d \operatorname{Tan}[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0$

1: $\int \frac{A + B \operatorname{Tan}[e + f x]}{\sqrt{a + b \operatorname{Tan}[e + f x]} \sqrt{c + d \operatorname{Tan}[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land A^2 + B^2 == 0$

Derivation: Integration by substitution

Basis: If $A^2 + B^2 == 0$, then $A + B \operatorname{Tan}[e + f x] == \frac{A^2}{f} \operatorname{Subst}\left[\frac{1}{A - B x}, x, \operatorname{Tan}[e + f x]\right] \partial_x \operatorname{Tan}[e + f x]$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land A^2 + B^2 == 0$, then

$$\int \frac{A + B \operatorname{Tan}[e + f x]}{\sqrt{a + b \operatorname{Tan}[e + f x]} \sqrt{c + d \operatorname{Tan}[e + f x]}} \, dx \to \frac{A^2}{f} \operatorname{Subst}\left[\int \frac{1}{(A - B x) \sqrt{a + b x} \sqrt{c + d x}} \, dx, x, \operatorname{Tan}[e + f x]\right]$$

Program code:

```
(* Int[(A_.+B_.*tan[e_.+f_.*x_])/(Sqrt[a_.+b_.*tan[e_.+f_.*x_]]*Sqrt[c_.+d_.*tan[e_.+f_.*x_]]),x_Symbol] :=
  A^2/f*Subst[Int[1/((A-B*x)*Sqrt[a+b*x]*Sqrt[c+d*x]),x],x,Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && EqQ[A^2+B^2,0] *)
```

2: $\int \frac{A+B\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\land a^2+b^2\neq 0\land c^2+d^2\neq 0\land A^2+B^2\neq 0$

Derivation: Algebraic expansion

Basis: $A+B\,z=\frac{A+i\,B}{2}\,(1-i\,z)+\frac{A-i\,B}{2}\,(1+i\,z)$

Rule: If $b\,c-a\,d\neq 0\land a^2+b^2\neq 0\land c^2+d^2\neq 0\land A^2+B^2\neq 0$, then

$$\int \frac{A+B\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx \to \frac{A+i\,B}{2}\int \frac{1-i\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx+\frac{A-i\,B}{2}\int \frac{1+i\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx$$

Program code:

```
(* Int[(A_.+B_.*tan[e_.+f_.*x_])/(Sqrt[a_.+b_.*tan[e_.+f_.*x_]]*Sqrt[c_.+d_.*tan[e_.+f_.*x_]]),x_Symbol] :=
  (A+I*B)/2*Int[(1-I*Tan[e+f*x])/(Sqrt[a+b*Tan[e+f*x]]*Sqrt[c+d*Tan[e+f*x]]),x] +
  (A-I*B)/2*Int[(1+I*Tan[e+f*x])/(Sqrt[a+b*Tan[e+f*x]]*Sqrt[c+d*Tan[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && NeQ[A^2+B^2,0] *)
```

7. $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0$

1: $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land A^2 + B^2 == 0$

Derivation: Integration by substitution

Basis: If $A^2 + B^2 == 0$, then $A + B \operatorname{Tan}[e + f x] == \frac{A^2}{f} \operatorname{Subst}\left[\frac{1}{A - B x}, x, \operatorname{Tan}[e + f x]\right] \partial_x \operatorname{Tan}[e + f x]$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land A^2 + B^2 == 0$, then

$$\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x]) \, dx \longrightarrow \frac{A^2}{f} \operatorname{Subst}\left[\int \frac{(a + b x)^m (c + d x)^n}{A - B x} \, dx, x, \operatorname{Tan}[e + f x]\right]$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  A^2/f*Subst[Int[(a+b*x)^m*(c+d*x)^n/(A-B*x),x],x,Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && EqQ[A^2+B^2,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(A+B\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\wedge a^2+b^2\neq 0\wedge A^2+B^2\neq 0$

Derivation: Algebraic expansion

Basis: $A+B\,z=\frac{A+i\,B}{2}\,(1-i\,z)+\frac{A-i\,B}{2}\,(1+i\,z)$

Rule: If $b\,c-a\,d\neq 0\wedge a^2+b^2\neq 0\wedge A^2+B^2\neq 0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{A+i\,B}{2}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(1-i\,\mathrm{Tan}[e+f\,x])\,dx+\frac{A-i\,B}{2}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(1+i\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (A+I*B)/2*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n*(1-I*Tan[e+f*x]),x] +
  (A-I*B)/2*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n*(1+I*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[A^2+B^2,0]
```