Rules for integrands of the form $(a+b\,\mathrm{Tan}[e+f\,x])^m\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)$

1: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\left(A+A\,\mathrm{Tan}[e+f\,x]^2\right)dx$

Derivation: Integration by substitution

Basis:
$F[b\,\mathrm{Tan}[e+f\,x]]\left(A+A\,\mathrm{Tan}[e+f\,x]^2\right)=\frac{A}{b\,f}\,\mathrm{Subst}[F[x],x,b\,\mathrm{Tan}[e+f\,x]]\,\partial_x\,(b\,\mathrm{Tan}[e+f\,x])$

Rule:

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\left(A+A\,\mathrm{Tan}[e+f\,x]^2\right)dx\,\longrightarrow\,\frac{A}{b\,f}\,\mathrm{Subst}\left[\int(a+x)^m\,dx,\,x,\,b\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(A_+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  A/(b*f)*Subst[Int[(a+x)^m,x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A,C]
```

```
Int[(a_.+b_.*cot[e_.+f_.*x_])^m_.*(A_+C_.*cot[e_.+f_.*x_]^2),x_Symbol] :=
  -A/(b*f)*Subst[Int[(a+x)^m,x],x,b*Cot[e+f*x]] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A,C]
```

2: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(A+B\,\text{Tan}[e+f\,x]+C\,\text{Tan}[e+f\,x]^2)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C==0$

Derivation: Algebraic simplification

Basis: If $A\,b^2-a\,b\,B+a^2\,C==0$, then $A+B\,z+C\,z^2==\frac{1}{b^2}\,(a+b\,z)\,(b\,B-a\,C+b\,C\,z)$

Rule: If $A\,b^2-a\,b\,B+a^2\,C==0$, then

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(A+B\,\text{Tan}[e+f\,x]+C\,\text{Tan}[e+f\,x]^2)\,dx\,\longrightarrow\,\frac{1}{b^2}\int (a+b\,\text{Tan}[e+f\,x])^{m+1}\,(b\,B-a\,C+b\,C\,\text{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  1/b^2*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[b*B-a*C+b*C*Tan[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && EqQ[A*b^2-a*b*B+a^2*C,0]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  -C/b^2*Int[(a+b*Tan[e+f*x])^(m+1)*(a-b*Tan[e+f*x]),x] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A*b^2+a^2*C,0]
```

3. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(A+B\,\text{Tan}[e+f\,x]+C\,\text{Tan}[e+f\,x]^2)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C\neq 0$

1. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(A+B\,\text{Tan}[e+f\,x]+C\,\text{Tan}[e+f\,x]^2)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\wedge\,m\leq -1$

1: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(A+B\,\text{Tan}[e+f\,x]+C\,\text{Tan}[e+f\,x]^2)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\wedge\,m\leq -1\,\wedge\,a^2+b^2==0$

Derivation: Algebraic expansion, symmetric tangent recurrence 2b with $m\to 0$ and symmetric tangent recurrence 2a with $A\to 0,\;B\to 1,\;m\to 1$

Rule: If $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\wedge\,m\leq -1\,\wedge\,a^2+b^2==0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m \left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right) dx \longrightarrow$$

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m (A+B\,\mathrm{Tan}[e+f\,x])\, dx + C\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,\mathrm{Tan}[e+f\,x]^2\, dx \longrightarrow$$

$$-\frac{(a\,A+b\,B-a\,C)\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Tan}[e+f\,x])^m}{2\,a\,f\,m}+$$

$$\frac{1}{2\,a^2\,m}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\left((b\,B-a\,C)+a\,A\,(2\,m+1)-(b\,C\,(m-1)+(A\,b-a\,B)\,(m+1))\,\mathrm{Tan}[e+f\,x]\right) dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  -(a*A+b*B-a*C)*Tan[e+f*x]*(a+b*Tan[e+f*x])^m/(2*a*f*m) +
  1/(2*a^2*m)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[(b*B-a*C)+a*A*(2*m+1)-(b*C*(m-1)+(A*b-a*B)*(m+1))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[A*b^2-a*b*B+a^2*C,0] && LeQ[m,-1] && EqQ[a^2+b^2,0]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  -(a*A-a*C)*Tan[e+f*x]*(a+b*Tan[e+f*x])^m/(2*a*f*m) +
  1/(2*a^2*m)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[-a*C+a*A*(2*m+1)-(b*C*(m-1)+A*b*(m+1))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && NeQ[A*b^2+a^2*C,0] && LeQ[m,-1] && EqQ[a^2+b^2,0]
```

2. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\land\,m\leq -1\,\land\,a^2+b^2\neq 0$

   1. $\int \frac{A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\land\,a^2+b^2\neq 0$

      1: $\int \frac{A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $a^2+b^2\neq 0\,\land\,A\,b-a\,B-b\,C=0$

Derivation: Algebraic expansion

Basis: If $A\,b-a\,B-b\,C=0$, then $\frac{A+B\,z+C\,z^2}{a+b\,z}=\frac{a\,A+b\,B-a\,C}{a^2+b^2}+\frac{(A\,b^2-a\,b\,B+a^2\,C)\,(1+z^2)}{(a^2+b^2)\,(a+b\,z)}$

Note: If $a^2+b^2\neq 0\,\land\,A\,b-a\,B-b\,C=0$, then $A\,b^2-a\,b\,B+a^2\,C\neq 0$.

Rule: If $a^2+b^2\neq 0\,\land\,A\,b-a\,B-b\,C=0$, then

$$\int \frac{A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx\,\longrightarrow\,\frac{(a\,A+b\,B-a\,C)\,x}{a^2+b^2}+\frac{A\,b^2-a\,b\,B+a^2\,C}{a^2+b^2}\int \frac{1+\mathrm{Tan}[e+f\,x]^2}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(A_+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2)/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (a*A+b*B-a*C)*x/(a^2+b^2) +
  (A*b^2-a*b*B+a^2*C)/(a^2+b^2)*Int[(1+Tan[e+f*x]^2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[a^2+b^2,0] && EqQ[A*b-a*B-b*C,0]
```

2. $\int \frac{A + B \operatorname{Tan}[e + f x] + C \operatorname{Tan}[e + f x]^2}{a + b \operatorname{Tan}[e + f x]} \, dx$ when $A b^2 - a b B + a^2 C \neq 0 \land a^2 + b^2 \neq 0 \land A b - a B - b C \neq 0$

1: $\int \frac{A + B \operatorname{Tan}[e + f x] + C \operatorname{Tan}[e + f x]^2}{\operatorname{Tan}[e + f x]} \, dx$ when $A - C \neq 0$

Derivation: Algebraic expansion

Rule: If $A - C \neq 0$, then

$$\int \frac{A + B \operatorname{Tan}[e + f x] + C \operatorname{Tan}[e + f x]^2}{\operatorname{Tan}[e + f x]} \, dx \longrightarrow B x + A \int \frac{1}{\operatorname{Tan}[e + f x]} \, dx + C \int \operatorname{Tan}[e + f x] \, dx$$

Program code:

```
Int[(A_+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2)/tan[e_.+f_.*x_],x_Symbol] :=
  B*x+A*Int[1/Tan[e+f*x],x] + C*Int[Tan[e+f*x],x] /;
FreeQ[{e,f,A,B,C},x] && NeQ[A,C]
```

```
Int[(A_+C_.*tan[e_.+f_.*x_]^2)/tan[e_.+f_.*x_],x_Symbol] :=
  A*Int[1/Tan[e+f*x],x] + C*Int[Tan[e+f*x],x] /;
FreeQ[{e,f,A,C},x] && NeQ[A,C]
```

2: $\int \frac{A + B \operatorname{Tan}[e + f x] + C \operatorname{Tan}[e + f x]^2}{a + b \operatorname{Tan}[e + f x]} \, dx$ when $A b^2 - a b B + a^2 C \neq 0 \land a^2 + b^2 \neq 0 \land A b - a B - b C \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{a+b z} == \frac{a A+b B-a C}{a^2+b^2} - \frac{(A b-a B-b C) z}{a^2+b^2} + \frac{(A b^2-a b B+a^2 C) (1+z^2)}{(a^2+b^2) (a+b z)}$

Rule: If $A b^2 - a b B + a^2 C \neq 0 \land a^2 + b^2 \neq 0 \land A b - a B - b C \neq 0$, then

$$\int \frac{A + B \operatorname{Tan}[e + f x] + C \operatorname{Tan}[e + f x]^2}{a + b \operatorname{Tan}[e + f x]} \, dx \longrightarrow \frac{(a A + b B - a C)\, x}{a^2 + b^2} - \frac{A b - a B - b C}{a^2 + b^2} \int \operatorname{Tan}[e + f x] \, dx + \frac{A b^2 - a b B + a^2 C}{a^2 + b^2} \int \frac{1 + \operatorname{Tan}[e + f x]^2}{a + b \operatorname{Tan}[e + f x]} \, dx$$

Program code:

```
Int[(A_+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2)/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (a*A+b*B-a*C)*x/(a^2+b^2) -
  (A*b-a*B-b*C)/(a^2+b^2)*Int[Tan[e+f*x],x] +
  (A*b^2-a*b*B+a^2*C)/(a^2+b^2)*Int[(1+Tan[e+f*x]^2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[A*b^2-a*b*B+a^2*C,0] && NeQ[a^2+b^2,0] && NeQ[A*b-a*B-b*C,0]
```

```
Int[(A_+C_.*tan[e_.+f_.*x_]^2)/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  a*(A-C)*x/(a^2+b^2) -
  b*(A-C)/(a^2+b^2)*Int[Tan[e+f*x],x] +
  (a^2*C+A*b^2)/(a^2+b^2)*Int[(1+Tan[e+f*x]^2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,e,f,A,C},x] && NeQ[a^2*C+A*b^2,0] && NeQ[a^2+b^2,0] && NeQ[A,C]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\land\,m<-1\,\land\,a^2+b^2\neq 0$

Derivation: Nondegenerate tangent recurrence 1a with $n\to 0,\ p\to 0$

Rule: If $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\land\,n<-1\,\land\,a^2+b^2\neq 0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)\,dx\,\longrightarrow$$

$$\frac{\left(A\,b^2-a\,b\,B+a^2\,C\right)\,(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}}{b\,f\,(m+1)\,\left(a^2+b^2\right)}+\frac{1}{a^2+b^2}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,\left(b\,B+a\,(A-C)-\left(A\,b-a\,B-b\,C\right)\,\mathrm{Tan}[e+f\,x]\right)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2-a*b*B+a^2*C)*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)*(a^2+b^2)) +
  1/(a^2+b^2)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[b*B+a*(A-C)-(A*b-a*B-b*C)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[A*b^2-a*b*B+a^2*C,0] && LtQ[m,-1] && NeQ[a^2+b^2,0]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2+a^2*C)*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)*(a^2+b^2)) +
  1/(a^2+b^2)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[a*(A-C)-(A*b-b*C)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && NeQ[A*b^2+a^2*C,0] && LtQ[m,-1] && NeQ[a^2+b^2,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\land\,m\nleq -1$

Derivation: Nondegenerate tangent recurrence 1b with $m\to 0,\ p\to 0$

Rule: If $A\,b^2-a\,b\,B+a^2\,C\neq 0\,\land\,m\nleq -1$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx\,\longrightarrow\,\frac{C\,(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}}{b\,f\,(m+1)}+\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(A-C+B\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  C*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)) + Int[(a+b*Tan[e+f*x])^m*Simp[A-C+B*Tan[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && NeQ[A*b^2-a*b*B+a^2*C,0] && Not[LeQ[m,-1]]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  C*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)) + (A-C)*Int[(a+b*Tan[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,A,C,m},x] && NeQ[A*b^2+a^2*C,0] && Not[LeQ[m,-1]]
```