1.
$$\int u (v + w)^p dx$$
 when $v == 0$
x: $\int u (v + w)^p dx$ when $v == 0$

Derivation: Algebraic simplification

Note: Many rules assume coefficients are not unrecognized zeros.

Note: Unfortunately this rule is commented out because it is too inefficient.

Rule: If v = 0, then

$$\int \!\! u \; \left(v + w \right)^p \, \mathrm{d} x \; \longrightarrow \; \int \!\! u \; w^p \, \mathrm{d} x$$

```
(* Int[u_.*(v_+w_)^p_.,x_Symbol] :=
   Int[u*w^p,x] /;
FreeQ[p,x] && EqQ[v,0] *)
```

1:
$$\int u (a + b x^n)^p dx \text{ when } a == 0$$

Derivation: Algebraic simplification

Rule: If a == 0, then

$$\int u \, \left(a + b \, \, x^n \right)^p \, \mathrm{d} \, x \, \, \longrightarrow \, \, \int u \, \left(b \, \, x^n \right)^p \, \mathrm{d} \, x$$

Program code:

```
Int[u_.*(a_+b_.*x_^n_.)^p_.,x_Symbol] :=
   Int[u*(b*x^n)^p,x] /;
FreeQ[{a,b,n,p},x] && EqQ[a,0]
```

2:
$$\int u (a + b x^n)^p dx$$
 when $b = 0$

Derivation: Algebraic simplification

Rule: If b == 0, then

$$\int u \, \left(a + b \, x^n\right)^p \, \mathrm{d}x \, \, \longrightarrow \, \, \int u \, \, a^p \, \, \mathrm{d}x$$

```
Int[u_.*(a_.+b_.*x_^n_.)^p_.,x_Symbol] :=
   Int[u*a^p,x] /;
FreeQ[{a,b,n,p},x] && EqQ[b,0]
```

3:
$$\int u (a + b x^n + c x^{2n})^p dx$$
 when $a == 0$

Derivation: Algebraic simplification

Rule: If a == 0, then

$$\int \! u \, \left(\, a \, + \, b \, \, x^n \, + \, c \, \, x^{2 \, n} \right)^{\, p} \, \mathbb{d} \, x \, \, \longrightarrow \, \, \int \! u \, \, \left(\, b \, \, x^n \, + \, c \, \, x^{2 \, n} \right)^{\, p} \, \mathbb{d} \, x$$

Program code:

```
Int[u_.*(a_+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
   Int[u*(b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[j,2*n] && EqQ[a,0]
```

4:
$$\int u (a + b x^n + c x^{2n})^p dx$$
 when $b == 0$

Derivation: Algebraic simplification

Rule: If b == 0, then

$$\int u \, \left(a + b \, \, x^n + c \, \, x^{2 \, n} \right)^p \, \mathrm{d}x \ \longrightarrow \ \int u \, \left(a + c \, \, x^{2 \, n} \right)^p \, \mathrm{d}x$$

```
Int[u_.*(a_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
   Int[u*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[j,2*n] && EqQ[b,0]
```

5:
$$\int u (a + b x^n + c x^{2n})^p dx$$
 when $c = 0$

Derivation: Algebraic simplification

Rule: If c == 0, then

$$\int \! u \, \left(a + b \, x^n + c \, x^{2\,n} \right)^p \, \text{d} x \ \longrightarrow \ \int \! u \, \left(a + b \, x^n \right)^p \, \text{d} x$$

Program code:

```
Int[u_.*(a_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
   Int[u*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[j,2*n] && EqQ[c,0]
```

2: $\int u (a v + b v + w)^p dx$ when v depends on x

Derivation: Algebraic simplification

Rule: If v depends on x, then

$$\int u \, \left(a \, v + b \, v + w \right)^p \, \text{d} \, x \ \longrightarrow \ \int u \, \left(\left(a + b \right) \, v + w \right)^p \, \text{d} \, x$$

```
Int[u_.*(a_.*v_+b_.*v_+w_.)^p_.,x_Symbol] :=
   Int[u*((a+b)*v+w)^p,x] /;
FreeQ[{a,b},x] && Not[FreeQ[v,x]]
```

3: $\int u P[x]^p dx$ when $p \notin \mathbb{Q} \wedge Simplify[p] \in \mathbb{Q}$

Derivation: Algebraic simplification

Note: Rubi's integration rules assume integer and rational exponents are recognized as such.

Rule: If $p \notin \mathbb{Q} \land Simplify[p] \in \mathbb{Q}$, then

$$\int\! u\,P[\,x\,]^{\,p}\,u\,\,\text{d}\,x\,\,\rightarrow\,\,\int\! u\,\,P[\,x\,]^{\,\text{Simplify}[\,p\,]}\,\,\text{d}\,x$$

Program code:

```
Int[u_.*Px_^p_,x_Symbol] :=
  Int[u*Px^Simplify[p],x] /;
PolyQ[Px,x] && Not[RationalQ[p]] && FreeQ[p,x] && RationalQ[Simplify[p]]
```

faudx
 fadx

Reference: CRC 1

Rule:

$$\int a \, dx \, \to \, a \, x$$

```
Int[a_,x_Symbol] :=
    a*x /;
FreeQ[a,x]
```

2:
$$\int a (b + c x) dx$$

Derivation: Power rule for integration

Rule:

$$\int a \, \left(b + c \, x \right) \, dx \, \, \rightarrow \, \, \frac{a \, \left(b + c \, x \right)^2}{2 \, c}$$

```
Int[a_*(b_+c_.*x_),x_Symbol] :=
  a*(b+c*x)^2/(2*c) /;
FreeQ[{a,b,c},x]
```

3: $\int a u dx$

Reference: G&R 2.02.1, CRC 2

Derivation: Constant extraction

Note: Since the rule for extracting the imaginary unit from integrands includes the function Identity, it is not displayed when showing steps thus avoiding trivial steps when integrating expressions involving hyperbolic functions.

Rule:

$$\int a \, u \, dx \, \rightarrow \, a \, \int u \, dx$$

```
Int[-u_,x_Symbol] :=
   Identity[-1]*Int[u,x]

Int[Complex[0,a_]*u_,x_Symbol] :=
   Complex[Identity[0],a]*Int[u,x] /;
FreeQ[a,x] && EqQ[a^2,1]

Int[a_*u_,x_Symbol] :=
   a*Int[u,x] /;
FreeQ[a,x] && Not[MatchQ[u, b_*v_ /; FreeQ[b,x]]]
```

```
5: \int a u + b v + \cdots dx
```

Reference: G&R 2.02.2, 2.111.1 CRC 2, 4, 23, 27

Note: By actually integrating linear power of x terms, this rule eliminates numerous trivial integration steps.

Rule:

$$\int a\; u\; +\; b\; v\; +\; \cdots\; \mathrm{d}\; x\; \; \longrightarrow \; a\; \int u\; \mathrm{d}\; x\; +\; b\; \int v\; \mathrm{d}\; x\; +\; \cdots$$

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
    ShowStep["","Int[a*u + b*v + ...,x]","a*Integrate[u,x] + b*Integrate[v,x] + ...",Hold[
    IntSum[u,x]]] /;
SimplifyFlag && SumQ[u],

Int[u_,x_Symbol] :=
    IntSum[u,x] /;
SumQ[u]]
```

6:
$$\int (c x)^m (u + v + \cdots) dx$$

Derivation: Algebraic expansion

Rule:

$$\int \left(c \; x \right)^m \; \left(u + v + \cdots \right) \; \mathrm{d} x \; \longrightarrow \; \int \left(c \; x \right)^m u + \left(c \; x \right)^m v + \cdots \; \mathrm{d} x$$

```
Int[(c_.*x_)^m_.*u_,x_Symbol] :=
   Int[ExpandIntegrand[(c*x)^m*u,x],x] /;
FreeQ[{c,m},x] && SumQ[u] && Not[LinearQ[u,x]] && Not[MatchQ[u,a_+b_.*v_ /; FreeQ[{a,b},x] && InverseFunctionQ[v]]]
```

7.
$$\int u (a v)^m (b v)^n \cdots dx$$

1:
$$\int u (a x^n)^m dx$$
 when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{(a x^n)^m}{x^m n} = 0$$

Rule: If $m \notin \mathbb{Z}$, then

$$\int u \left(a x^{n}\right)^{m} x \rightarrow \frac{a^{\text{IntPart}[m]} \left(a x^{n}\right)^{\text{FracPart}[m]}}{x^{n \, \text{FracPart}[m]}} \int u \, x^{m \, n} \, dx$$

```
Int[u_.*(a_.*x_^n_)^m_,x_Symbol] :=
   a^IntPart[m]*(a*x^n)^FracPart[m]/x^(n*FracPart[m])*Int[u*x^(m*n),x] /;
FreeQ[{a,m,n},x] && Not[IntegerQ[m]]
```

2:
$$\int u v^m (b v)^n dx$$
 when $m \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $m \in \mathbb{Z}$, then $v^m = \frac{1}{b^m} (b v)^m$

Rule: If $m \in \mathbb{Z}$, then

$$\int\! u\; v^m\; \big(b\; v\big)^n\; \text{d} x\; \longrightarrow\; \frac{1}{b^m}\; \int\! u\; \big(b\; v\big)^{m+n}\; \text{d} x$$

```
Int[u_.*v_^m_.*(b_*v_)^n_,x_Symbol] :=
    1/b^m*Int[u*(b*v)^(m+n),x] /;
FreeQ[{b,n},x] && IntegerQ[m]
```

3. $\left[u \left(a v \right)^m \left(b v \right)^n dx \text{ when } m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \right]$

1.
$$\int u \, \left(a \, v\right)^m \, \left(b \, v\right)^n \, \mathrm{d}x \text{ when } m \notin \mathbb{Z} \ \land \ n + \frac{1}{2} \in \mathbb{Z}$$

1.
$$\int u (a v)^m (b v)^n dx$$
 when $m \notin \mathbb{Z} \wedge n + \frac{1}{2} \in \mathbb{Z}^+$

1:
$$\int u \ (a \ v)^m \ \left(b \ v\right)^n \, \mathrm{d}x \ \text{ when } m \notin \mathbb{Z} \ \land \ n+\frac{1}{2} \in \mathbb{Z}^+ \land \ m+n \in \mathbb{Z}$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{\sqrt{b F[x]}}{\sqrt{a F[x]}} = 0$$

Basis: If
$$n + \frac{1}{2} \in \mathbb{Z}$$
, then $(b \ v)^n = \frac{b^{n - \frac{1}{2}} \sqrt{b \ v}}{a^{n - \frac{1}{2}} \sqrt{a \ v}} (a \ v)^n$

Rule: If $m \notin \mathbb{Z} \wedge n + \frac{1}{2} \in \mathbb{Z}^+ \wedge m + n \in \mathbb{Z}$, then

$$\int u (a v)^m (b v)^n dx \rightarrow \frac{a^{m+\frac{1}{2}} b^{n-\frac{1}{2}} \sqrt{b v}}{\sqrt{a v}} \int u v^{m+n} dx$$

Program code:

X:
$$\int u \ (a \ v)^m \ (b \ v)^n \ dx \text{ when } m \notin \mathbb{Z} \ \land \ n + \frac{1}{2} \in \mathbb{Z}^+ \land \ m + n \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{\sqrt{b F[x]}}{\sqrt{a F[x]}} = 0$$

Basis: If
$$n + \frac{1}{2} \in \mathbb{Z}$$
, then $(b \ v)^n = \frac{b^{n - \frac{1}{2}} \sqrt{b \ v}}{a^{n - \frac{1}{2}} \sqrt{a \ v}} (a \ v)^n$

Rule: If $m \notin \mathbb{Z} \wedge n + \frac{1}{2} \in \mathbb{Z}^+ \wedge m + n \notin \mathbb{Z}$, then

$$\int u (av)^{m} (bv)^{n} dx \rightarrow \frac{b^{n-\frac{1}{2}} \sqrt{bv}}{a^{n-\frac{1}{2}} \sqrt{av}} \int u (av)^{m+n} dx$$

```
(* Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
   b^(n-1/2)*Sqrt[b*v]/(a^(n-1/2)*Sqrt[a*v])*Int[u*(a*v)^(m+n),x] /;
FreeQ[{a,b,m},x] && Not[IntegerQ[m]] && IGtQ[n+1/2,0] && Not[IntegerQ[m+n]] *)
```

2.
$$\int u \ (a \ v)^m \ (b \ v)^n \ dx \ \text{ when } m \notin \mathbb{Z} \ \wedge \ n - \frac{1}{2} \in \mathbb{Z}^-$$

$$1: \int u \ (a \ v)^m \ (b \ v)^n \ dx \ \text{ when } m \notin \mathbb{Z} \ \wedge \ n - \frac{1}{2} \in \mathbb{Z}^- \wedge \ m + n \in \mathbb{Z}$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{\sqrt{a F[x]}}{\sqrt{b F[x]}} = 0$$

Basis: If
$$n - \frac{1}{2} \in \mathbb{Z}$$
, then $(b \ v)^n = \frac{b^{n + \frac{1}{2}} \sqrt{a \ v}}{a^{n + \frac{1}{2}} \sqrt{b \ v}} (a \ v)^n$

Rule: If $m \notin \mathbb{Z} \wedge n - \frac{1}{2} \in \mathbb{Z}^- \wedge m + n \in \mathbb{Z}$, then

$$\int u (a v)^{m} (b v)^{n} dx \rightarrow \frac{a^{m-\frac{1}{2}} b^{n+\frac{1}{2}} \sqrt{a v}}{\sqrt{b v}} \int u v^{m+n} dx$$

```
Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
    a^(m-1/2)*b^(n+1/2)*Sqrt[a*v]/Sqrt[b*v]*Int[u*v^(m+n),x] /;
FreeQ[{a,b,m},x] && Not[IntegerQ[m]] && ILtQ[n-1/2,0] && IntegerQ[m+n]
```

X:
$$\int u \ (a \ v)^m \ (b \ v)^n \ dx \text{ when } m \notin \mathbb{Z} \ \land \ n - \frac{1}{2} \in \mathbb{Z}^- \land \ m + n \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{\sqrt{a F[x]}}{\sqrt{b F[x]}} = 0$$

Basis: If
$$n - \frac{1}{2} \in \mathbb{Z}$$
, then $(b \ v)^n = \frac{b^{n + \frac{1}{2}} \sqrt{a \ v}}{a^{n + \frac{1}{2}} \sqrt{b \ v}} (a \ v)^n$

Rule: If $m \notin \mathbb{Z} \ \land \ n - \frac{1}{2} \in \mathbb{Z}^- \land \ m + n \notin \mathbb{Z}$, then

$$\int u (av)^{m} (bv)^{n} dx \rightarrow \frac{b^{n+\frac{1}{2}} \sqrt{av}}{a^{n+\frac{1}{2}} \sqrt{bv}} \int u (av)^{m+n} dx$$

```
(* Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
  b^(n+1/2)*Sqrt[a*v]/(a^(n+1/2)*Sqrt[b*v])*Int[u*(a*v)^(m+n),x] /;
FreeQ[{a,b,m},x] && Not[IntegerQ[m]] && ILtQ[n-1/2,0] && Not[IntegerQ[m+n]] *)
```

2.
$$\int u \ (a \ v)^m \ (b \ v)^n \ dx \ \text{when} \ m \notin \mathbb{Z} \ \land \ n \notin \mathbb{Z}$$

$$1: \int u \ (a \ v)^m \ (b \ v)^n \ dx \ \text{when} \ m \notin \mathbb{Z} \ \land \ n \notin \mathbb{Z} \ \land \ m+n \in \mathbb{Z}$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{(b F[x])^n}{(a F[x])^n} = 0$$

Rule: If m $\notin \mathbb{Z} \ \land \ n \notin \mathbb{Z} \ \land \ m+n \in \mathbb{Z}$, then

$$\int u \, (a \, v)^m \, \left(b \, v\right)^n \, \mathrm{d}x \, \longrightarrow \, \frac{a^{m+n} \, \left(b \, v\right)^n}{\left(a \, v\right)^n} \int u \, v^{m+n} \, \mathrm{d}x$$

```
Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
    a^(m+n)*(b*v)^n/(a*v)^n*Int[u*v^(m+n),x] /;
FreeQ[{a,b,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[m+n]
```

2:
$$\int u \ (a \ v)^m \ (b \ v)^n \ dx \ \text{when} \ m \notin \mathbb{Z} \ \land \ n \notin \mathbb{Z} \ \land \ m+n \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{(b F[x])^n}{(a F[x])^n} = 0$$

Rule: If $m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land m + n \notin \mathbb{Z}$, then

$$\int u \, \left(a \, v\right)^{\, m} \, \left(b \, v\right)^{\, n} \, \mathrm{d}x \, \, \rightarrow \, \frac{b^{\, \text{IntPart}[\, n]} \, \left(b \, v\right)^{\, \text{FracPart}[\, n]}}{a^{\, \text{IntPart}[\, n]} \, \left(a \, v\right)^{\, \text{FracPart}[\, n]}} \, \int u \, \left(a \, v\right)^{\, m+n} \, \mathrm{d}x$$

```
Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
   b^IntPart[n]*(b*v)^FracPart[n]/(a^IntPart[n]*(a*v)^FracPart[n])*Int[u*(a*v)^(m+n),x] /;
FreeQ[{a,b,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[m+n]]
```

8.
$$\int u (a + b v)^m (c + d v)^n dx$$
 when $b c - a d == 0$

1: $\int u (a + b v)^m (c + d v)^n dx$ when $b c - a d == 0 \land (m \in \mathbb{Z} \lor \frac{b}{d} > 0)$

Derivation: Algebraic simplification

Basis: If
$$b c - a d = 0 \land (m \in \mathbb{Z} \lor \frac{b}{d} > 0)$$
, then $(a + b z)^m = (\frac{b}{d})^m (c + d z)^m$

Rule: If
$$b \ c - a \ d == 0 \ \land \ \left(m \in \mathbb{Z} \ \lor \ \frac{b}{d} > 0\right)$$
, then

$$\int u \left(a+b\,v\right)^m \left(c+d\,v\right)^n \,\mathrm{d}x \,\, \to \, \left(\frac{b}{d}\right)^m \int u \, \left(c+d\,v\right)^{m+n} \,\mathrm{d}x$$

Program code:

2:
$$\int u \left(a+b \ v\right)^m \left(c+d \ v\right)^n \, \mathrm{d}x \text{ when } b \ c-a \ d == 0 \ \land \ \neg \ \left(m \in \mathbb{Z} \ \lor \ n \in \mathbb{Z} \ \lor \ \frac{b}{d} > 0\right)$$

Derivation: Piecewise constant extraction

Basis: If
$$b c - a d = 0$$
, then $\partial_x \frac{(a+b F[x])^n}{(c+d F[x])^n} = 0$

Rule: If
$$b \ c - a \ d == 0 \ \land \ \neg \ \left(m \in \mathbb{Z} \ \lor \ n \in \mathbb{Z} \ \lor \ \frac{b}{d} > 0 \right)$$
, then

$$\int u \left(a + b v\right)^{m} \left(c + d v\right)^{n} dx \rightarrow \frac{\left(a + b v\right)^{m}}{\left(c + d v\right)^{m}} \int u \left(c + d v\right)^{m+n} dx$$

Program code:

```
Int[u_.*(a_+b_.*v_)^m_*(c_+d_.*v_)^n_,x_Symbol] :=
   (a+b*v)^m/(c+d*v)^m*Int[u*(c+d*v)^(m+n),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[b*c-a*d,0] && Not[IntegerQ[m] || IntegerQ[n] || GtQ[b/d,0]]
```

```
9: \int u (a + b v)^m (A + B v + C v^2) dx when A b^2 - a b B + a^2 C == 0 \land m \le -1 ?? ??
```

Derivation: Algebraic simplification

Basis: If A
$$b^2 - ab B + a^2 C = 0$$
, then A + B z + C $z^2 = \frac{1}{b^2} (a + bz) (b B - a C + b C z)$

Rule: If A b^2 – a b B + a^2 C == 0 \land m \le –1, then

$$\int u \; \left(a\; v\right)^{\;m} \; \left(b\; v + c\; v^2\right) \; \mathrm{d}x \; \longrightarrow \; \frac{1}{a} \int u \; \left(a\; v\right)^{\;m+1} \; \left(b + c\; v\right) \; \mathrm{d}x$$

$$\int u \; \left(a + b\; v\right)^{\;m} \; \left(A + B\; v + C\; v^2\right) \; \mathrm{d}x \; \longrightarrow \; \frac{1}{b^2} \int u \; \left(a + b\; v\right)^{\;m+1} \; \left(b\; B - a\; C + b\; C\; v\right) \; \mathrm{d}x$$

```
(* Int[u_.*(a_.*v_)^m_*(b_.*v_+c_.*v_^2),x_Symbol] :=
    1/a*Int[u*(a*v)^(m+1)*(b+c*v),x] /;
FreeQ[{a,b,c},x] && LeQ[m,-1] *)
```

```
Int[u_.*(a_+b_.*v_)^m_*(A_.+B_.*v_+C_.*v_^2),x_Symbol] :=
    1/b^2*Int[u*(a+b*v)^(m+1)*Simp[b*B-a*C+b*C*v,x],x] /;
FreeQ[{a,b,A,B,C},x] && EqQ[A*b^2-a*b*B+a^2*C,0] && LeQ[m,-1]
```

10:
$$\int u (a + b x^n)^m (c + d x^{-n})^p dx$$
 when $ac - bd == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If
$$a \ c - b \ d == 0 \ \land \ p \in \mathbb{Z}$$
, then $(c + d \ x^{-n})^p = \left(\frac{d}{a}\right)^p \frac{(a+b \ x^n)^p}{x^{n \ p}}$

Rule: If a c – b d == $0 \land p \in \mathbb{Z}$, then

$$\int u \left(a+b \ x^n\right)^m \left(c+d \ x^{-n}\right)^p \, \mathrm{d}x \ \longrightarrow \left(\frac{d}{a}\right)^p \int \frac{u \left(a+b \ x^n\right)^{m+p}}{x^{n\,p}} \, \mathrm{d}x$$

```
Int[u_.*(a_+b_.*x_^n_.)^m_.*(c_+d_.*x_^q_.)^p_.,x_Symbol] :=
   (d/a)^p*Int[u*(a+b*x^n)^(m+p)/x^(n*p),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[q,-n] && IntegerQ[p] && EqQ[a*c-b*d,0] && Not[IntegerQ[m] && NegQ[n]]
```

11:
$$\int u (a + b x^n)^m (c + d x^{2n})^{-m} dx$$
 when $b^2 c + a^2 d == 0 \land a > 0 \land d < 0$

Derivation: Algebraic simplification

Basis: If
$$b^2 c + a^2 d == 0 \land a > 0 \land d < 0$$
, then $(a + b z)^m (c + d z^2)^{-m} = (-\frac{b^2}{d})^m (a - b z)^{-m}$

Rule: If $b^2 c + a^2 d = 0 \land a > 0 \land d < 0$, then

$$\int \! u \, \left(a + b \, x^n\right)^m \, \left(c + d \, x^{2\,n}\right)^{-m} \, \mathrm{d} x \ \longrightarrow \left(-\, \frac{b^2}{d}\right)^m \, \int \! u \, \left(a - b \, x^n\right)^{-m} \, \mathrm{d} x$$

```
Int[u_.*(a_+b_.*x_^n_.)^m_.*(c_+d_.*x_^j_)^p_.,x_Symbol] :=
   (-b^2/d)^m*Int[u*(a-b*x^n)^(-m),x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[j,2*n] && EqQ[p,-m] && EqQ[b^2*c+a^2*d,0] && GtQ[a,0] && LtQ[d,0]
```

12:
$$\int u (a + b x^n + c x^{2n})^p dx$$
 when $b^2 - 4 a c = 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If
$$b^2 - 4$$
 a c = 0, then $a + b z + c z^2 = \frac{1}{c} \left(\frac{b}{2} + c z \right)^2$

Basis: If
$$b^2 - 4$$
 a c == 0, then $a + b z + c z^2 = \left(\sqrt{a} + \frac{b z}{2\sqrt{a}}\right)^2$

Rule: If $b^2 - 4$ a $c = 0 \land p \in \mathbb{Z}$, then

$$\int u \, \left(a + b \, x^n + c \, x^{2\,n}\right)^p \, \mathrm{d}x \ \longrightarrow \ \frac{1}{c^p} \int u \, \left(\frac{b}{2} + c \, x^n\right)^{2\,p} \, \mathrm{d}x$$

```
Int[u_{*}(a_{+}b_{*}x_{+}c_{*}x_{^{2}})^{p}_{*},x_{Symbol}] := Int[u_{*}Cancel[(b/2+c_{*}x)^{(2*p)/c^{p}},x] /; FreeQ[\{a,b,c\},x] \&\& EqQ[b^{2}-4*a*c,0] \&\& IntegerQ[p]
```

```
Int[u_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
    1/c^p*Int[u*(b/2+c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```