### Rules for integrands of the form $(a + b Sec [c + d x^n])^p$

## Derivation: Integration by substitution

Basis: If 
$$-1 \le n \le 1 \land n \ne 0$$
, then  $F[x^n] = \frac{1}{n} \operatorname{Subst} \left[ x^{\frac{1}{n-1}} F[x], x, x^n \right] \partial_x x^n$ 

Note: If  $\frac{1}{n} \in \mathbb{Z}^-$ , resulting integrand is not integrable.

Rule: If 
$$\frac{1}{n} \in \mathbb{Z}^+ \land p \in \mathbb{Z}$$
, then

$$\int \left(a+b\,\text{Sec}\big[\,c+d\,x^n\big]\right)^p\,\text{d}x \ \to \ \frac{1}{n}\,\text{Subst}\Big[\int x^{\frac{1}{n}-1}\,\left(a+b\,\text{Sec}\big[\,c+d\,x\big]\right)^p\,\text{d}x\,,\,\,x\,,\,\,x^n\Big]$$

```
Int[(a_.+b_.*Sec[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
    1/n*Subst[Int[x^(1/n-1)*(a+b*Sec[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]

Int[(a_.+b_.*Csc[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
    1/n*Subst[Int[x^(1/n-1)*(a+b*Csc[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]
```

X: 
$$\int (a + b \operatorname{Sec}[c + d x^n])^p dx$$

Rule:

$$\Big\lceil \left( \mathsf{a} + \mathsf{b} \, \mathsf{Sec} \big[ \mathsf{c} + \mathsf{d} \, \mathsf{x}^{\mathsf{n}} \big] \right)^{\mathsf{p}} \, \mathbb{d} \mathsf{x} \,\, \longrightarrow \,\, \Big\lceil \left( \mathsf{a} + \mathsf{b} \, \mathsf{Sec} \big[ \mathsf{c} + \mathsf{d} \, \mathsf{x}^{\mathsf{n}} \big] \right)^{\mathsf{p}} \, \mathbb{d} \mathsf{x}$$

## Program code:

```
Int[(a_.+b_.*Sec[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
   Unintegrable[(a+b*Sec[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]

Int[(a_.+b_.*Csc[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
   Unintegrable[(a+b*Csc[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]
```

S:  $\int (a + b \operatorname{Sec}[c + d u^n])^p dx$  when u == e + f x

Derivation: Integration by substitution

Rule: If u == e + f x, then

$$\int \left(a + b \operatorname{Sec}\left[c + d u^{n}\right]\right)^{p} dx \ \to \ \frac{1}{f} \operatorname{Subst}\left[\int \left(a + b \operatorname{Sec}\left[c + d x^{n}\right]\right)^{p} dx, \ x, \ u\right]$$

```
Int[(a_.+b_.*Sec[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
    1/Coefficient[u,x,1]*Subst[Int[(a+b*Sec[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(a_.+b_.*Csc[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
    1/Coefficient[u,x,1]*Subst[Int[(a+b*Csc[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

N: 
$$\int (a + b \operatorname{Sec}[u])^{p} dx \text{ when } u = c + dx^{n}$$

Derivation: Algebraic normalization

Rule: If 
$$u = c + d x^n$$
, then

$$\int \left(a+b \, \mathsf{Sec}\,[u]\right)^p \, \mathrm{d}x \ \to \ \int \left(a+b \, \mathsf{Sec}\, \big[c+d \, x^n\big]\right)^p \, \mathrm{d}x$$

```
Int[(a_.+b_.*Sec[u_])^p_.,x_Symbol] :=
    Int[(a+b*Sec[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]

Int[(a_.+b_.*Csc[u_])^p_.,x_Symbol] :=
    Int[(a+b*Csc[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

#### Rules for integrands of the form $(e x)^m (a + b Sec[c + d x^n])^p$

1. 
$$\int x^m \left(a + b \operatorname{Sec}\left[c + d \ x^n\right]\right)^p dx$$
 
$$1: \int x^m \left(a + b \operatorname{Sec}\left[c + d \ x^n\right]\right)^p dx \text{ when } \frac{m+1}{n} \in \mathbb{Z}^+ \wedge \ p \in \mathbb{Z}$$

Derivation: Integration by substitution

Basis: If 
$$\frac{m+1}{n} \in \mathbb{Z}$$
, then  $x^m \, F[x^n] = \frac{1}{n} \, \text{Subst} \big[ x^{\frac{m+1}{n}-1} \, F[x] \,, \, x \,, \, x^n \big] \, \partial_x \, x^n$ 

Note: If  $\frac{m+1}{n} \in \mathbb{Z}^-$ , resulting integrand is not integrable.

Rule: If  $\frac{m+1}{n} \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}$ , then

$$\int \! x^m \, \left(a + b \, \text{Sec} \big[c + d \, x^n\big]\right)^p \, \text{d}x \,\, \rightarrow \,\, \frac{1}{n} \, \text{Subst} \Big[\int \! x^{\frac{n+1}{n}-1} \, \left(a + b \, \text{Sec} \big[c + d \, x\big]\right)^p \, \text{d}x \,, \,\, x \,, \,\, x^n\Big]$$

```
Int[x_^m_.*(a_.+b_.*Sec[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
    1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Sec[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]

Int[x_^m_.*(a_.+b_.*Csc[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
    1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Csc[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]
```

X: 
$$\int x^m (a + b Sec[c + d x^n])^p dx$$

Rule:

$$\int \! x^m \, \left( a + b \, \text{Sec} \left[ \, c + d \, \, x^n \, \right] \right)^p \, \text{d} \, x \, \, \rightarrow \, \, \int \! x^m \, \left( a + b \, \, \text{Sec} \left[ \, c + d \, \, x^n \, \right] \right)^p \, \text{d} \, x$$

```
Int[x_^m_.*(a_.+b_.*Sec[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
   Unintegrable[x^m*(a+b*Sec[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]

Int[x_^m_.*(a_.+b_.*Csc[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
   Unintegrable[x^m*(a+b*Csc[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

2: 
$$\int (e x)^m (a + b Sec[c + d x^n])^p dx$$

Derivation: Piecewise constant extraction

Basis: 
$$\partial_x \frac{(e \times)^m}{x^m} = 0$$

Rule:

$$\int \left(e\;x\right)^{m} \, \left(a+b\; Sec\left[c+d\;x^{n}\right]\right)^{p} \, \mathrm{d}x \; \to \; \frac{e^{\text{IntPart}[m]} \, \left(e\;x\right)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int \! x^{m} \, \left(a+b\; Sec\left[c+d\;x^{n}\right]\right)^{p} \, \mathrm{d}x$$

```
Int[(e_*x_)^m_.*(a_.+b_.*Sec[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
    e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sec[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]

Int[(e_*x_)^m_.*(a_.+b_.*Csc[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
    e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Csc[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

N: 
$$\int (e x)^m (a + b Sec[u])^p dx when u == c + d x^n$$

Derivation: Algebraic normalization

Rule: If 
$$u = c + d x^n$$
, then

$$\int \left(e\,x\right)^{\,m}\,\left(a+b\,Sec\left[u\right]\right)^{\,p}\,\mathrm{d}x\ \longrightarrow\ \int \left(e\,x\right)^{\,m}\,\left(a+b\,Sec\left[\,c+d\,\,x^{n}\right]\right)^{\,p}\,\mathrm{d}x$$

```
Int[(e_*x_)^m_.*(a_.+b_.*Sec[u_])^p_.,x_Symbol] :=
    Int[(e*x)^m*(a+b*Sec[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]

Int[(e_*x_)^m_.*(a_.+b_.*Csc[u_])^p_.,x_Symbol] :=
    Int[(e*x)^m*(a+b*Csc[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

### Rules for integrands of the form $x^m Sec[a + b x^n]^p Sin[a + b x^n]$

1: 
$$\left[x^{m} \operatorname{Sec}\left[a + b \ x^{n}\right]^{p} \operatorname{Sin}\left[a + b \ x^{n}\right] dx \text{ when } n \in \mathbb{Z} \land m - n \ge 0 \land p \ne 1\right]$$

# Derivation: Integration by parts

Rule: If  $n \in \mathbb{Z} \land m - n \ge 0 \land p \ne 1$ , then

$$\int x^m \, \mathsf{Sec} \big[ a + b \, x^n \big]^p \, \mathsf{Sin} \big[ a + b \, x^n \big] \, \mathrm{d} x \, \longrightarrow \, \frac{x^{m-n+1} \, \mathsf{Sec} \big[ a + b \, x^n \big]^{p-1}}{b \, n \, (p-1)} - \frac{m-n+1}{b \, n \, (p-1)} \, \int x^{m-n} \, \mathsf{Sec} \big[ a + b \, x^n \big]^{p-1} \, \mathrm{d} x$$

```
Int[x_^m_.*Sec[a_.+b_.*x_^n_.]^p_*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
    x^(m-n+1)*Sec[a+b*x^n]^(p-1)/(b*n*(p-1)) -
    (m-n+1)/(b*n*(p-1))*Int[x^(m-n)*Sec[a+b*x^n]^(p-1),x] /;
FreeQ[{a,b,p},x] && IntegerQ[n] && GeQ[m-n,0] && NeQ[p,1]

Int[x_^m_.*Csc[a_.+b_.*x_^n_.]^p_*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
    -x^(m-n+1)*Csc[a+b*x^n]^(p-1)/(b*n*(p-1)) +
    (m-n+1)/(b*n*(p-1))*Int[x^(m-n)*Csc[a+b*x^n]^(p-1),x] /;
FreeQ[{a,b,p},x] && IntegerQ[n] && GeQ[m-n,0] && NeQ[p,1]
```