Rules for integrands of the form $u \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s$

1: $\int u \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $b c-a d=0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c-a d=0$, then $a+b x=\frac{b}{d}(c+d x)$

Rule: If $b c-a d=0 \land p \in \mathbb{Z}$, then

$$\int u \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s dx \to \int u \operatorname{Log}\left[e\left(\frac{b^p f}{d^p}(c+d x)^{p+q}\right)^r\right]^s dx$$

Program code:

```
Int[u_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(b^p*f/d^p*(c+d*x)^(p+q))^r]^s,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && EqQ[b*c-a*d,0] && IntegerQ[p]
```

2. $\int \text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s dx$ when $b\,c-a\,d\neq 0$

1: $\int \text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s dx$ when $b\,c-a\,d\neq 0\land p+q=0\land s\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $1=\partial_x\frac{a+b\,x}{b}$

Basis: If $p+q=0$, then $\partial_x \text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s=-\frac{q\,r\,s\,(b\,c-a\,d)\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}$

Rule: If $b\,c-a\,d\neq 0\land p+q=0\land s\in\mathbb{Z}^+$, then

$$\int \text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s dx\to$$
$$\frac{(a+b\,x)\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{b}+\frac{q\,r\,s\,(b\,c-a\,d)}{b}\int\frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{c+d\,x}dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (a+b*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/b +
  q*r*s*(b*c-a*d)/b*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && IGtQ[s,0]
```

2: $\int \text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s dx$ when $b\,c-a\,d\neq 0\land p+q\neq 0\land s\in\mathbb{Z}^+\land s<4$

Derivation: Integration by parts

Basis: $1=\partial_x\frac{a+b\,x}{b}$

Basis: $\partial_x \operatorname{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s =$

$$\frac{b\,r\,s\,(p+q)\operatorname{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{a+b\,x}-\frac{q\,r\,s\,(b\,c-a\,d)\operatorname{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}$$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,p+q\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,s<4$, then

$$\int \operatorname{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx\,\longrightarrow$$

$$\frac{(a+b\,x)\operatorname{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{b}-$$
$$r\,s\,(p+q)\int \operatorname{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}dx+\frac{q\,r\,s\,(b\,c-a\,d)}{b}\int \frac{\operatorname{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{c+d\,x}\,dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (a+b*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/b -
  r*s*(p+q)*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1),x] +
  q*r*s*(b*c-a*d)/b*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && NeQ[p+q,0] && IGtQ[s,0] && LtQ[s,4]
```

3. $\int (g+h x)^m \operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0$

1. $\int (g+h x)^m \operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0 \land p+q == 0$

1. $\int \frac{\operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^s}{g+h x} dx$ when $b c-a d \neq 0 \land p+q == 0 \land s \in \mathbb{Z}^+$

1: $\int \frac{\operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^s}{g+h x} dx$ when $b c-a d \neq 0 \land p+q == 0 \land b g-a h == 0 \land s \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: If $b g-a h == 0$, then $\frac{1}{g+h x} == -\frac{1}{h} \partial_x \operatorname{Log}\left[-\frac{b c-a d}{d(a+b x)}\right]$

Basis: If $p+q == 0$, then $\partial_x \operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^s == \frac{p r s(b c-a d)}{(a+b x)(c+d x)} \operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d \neq 0 \land p+q == 0 \land b g-a h == 0 \land s \in \mathbb{Z}^+$, then

$$\int \frac{\operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^s}{g+h x} dx \rightarrow$$

$$-\frac{\operatorname{Log}\left[-\frac{b c-a d}{d(a+b x)}\right] \operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^s}{h}+\frac{p r s(b c-a d)}{h} \int \frac{\operatorname{Log}\left[-\frac{b c-a d}{d(a+b x)}\right] \operatorname{Log}\left[e\left(f(a+b x)^p(c+d x)^q\right)^r\right]^{s-1}}{(a+b x)(c+d x)} dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_./(g_.+h_.*x_),x_Symbol] :=
  -Log[-(b*c-a*d)/(d*(a+b*x))]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/h +
  p*r*s(b*c-a*d)/h*Int[Log[-(b*c-a*d)/(d*(a+b*x))]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[b*g-a*h,0] && IGtQ[s,0]
```

2: $\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{g+h x} \, dx$ when $b c-a d\neq 0\land p+q=0\land b g-a h\neq 0\land d g-c h\neq 0\land s\in \mathbb{Z}^+ +1$

Derivation: Algebraic expansion

Basis: $\frac{1}{g+h x}=\frac{d}{h (c+d x)}-\frac{d g-c h}{h (c+d x) (g+h x)}$

Rule: If $b c-a d\neq 0\land p+q=0\land b g-a h\neq 0\land d g-c h\neq 0\land s\in \mathbb{Z}^+ +1$, then

$$\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{g+h x} \, dx \longrightarrow$$
$$\frac{d}{h}\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{c+d x} \, dx-\frac{d g-c h}{h}\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(c+d x) (g+h x)} \, dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_/(g_.+h_.*x_),x_Symbol] :=
  d/h*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(c+d*x),x] -
  (d*g-c*h)/h*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/((c+d*x)*(g+h*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && NeQ[b*g-a*h,0] && NeQ[d*g-c*h,0] && IGtQ[s,1]
```

2: $\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(g+h x)^2} \, dx$ when $b c-a d\neq 0\land p+q=0\land b g-a h\neq 0\land s\in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\frac{1}{(g+h x)^2}=\partial_x \frac{a+b x}{(b g-a h) (g+h x)}$

Basis: If $p+q=0$, then $\partial_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s=\frac{p r s (b c-a d)}{(a+b x) (c+d x)} \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d\neq 0\land p+q=0\land b g-a h\neq 0\land s\in \mathbb{Z}^+$, then

$$\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(g+h x)^2} \, dx \longrightarrow$$

$$\frac{(a+b x) \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(b g-a h)(g+h x)}-\frac{p r s (b c-a d)}{(b g-a h)} \int \frac{\operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{(c+d x)(g+h x)} dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_./(g_.+h_.*x_)^2,x_Symbol] :=
  (a+b*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/((b*g-a*h)*(g+h*x)) -
  p*r*s*(b*c-a*d)/(b*g-a*h)*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((c+d*x)*(g+h*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && NeQ[b*g-a*h,0] && IGtQ[s,0]
```

3: $\int \frac{\operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(g+h x)^3} dx$ when $b c-a d \neq 0 \land p+q=0 \land b g-a h=0 \land d g-c h \neq 0 \land s \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{1}{(g+h x)^3}=\frac{d}{(d g-c h)(g+h x)^2}-\frac{h (c+d x)}{(d g-c h)(g+h x)^3}$

Rule: If $b c-a d \neq 0 \land p+q=0 \land b g-a h=0 \land d g-c h \neq 0 \land s \in \mathbb{Z}^+$, then

$$\int \frac{\operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(g+h x)^3} dx \longrightarrow$$

$$\frac{d}{d g-c h} \int \frac{\operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(g+h x)^2} dx-\frac{h}{d g-c h} \int \frac{(c+d x) \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(g+h x)^3} dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_/(g_.+h_.*x_)^3,x_Symbol] :=
  d/(d*g-c*h)*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(g+h*x)^2,x] -
  h/(d*g-c*h)*Int[(c+d*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(g+h*x)^3,x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[b*g-a*h,0] && NeQ[d*g-c*h,0] && IGtQ[s,0]
```

4: $\int (g+h\,x)^m\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,p+q==0\,\wedge\,s\in\mathbb{Z}^+\,\wedge\,m\neq -1$

Derivation: Integration by parts

Basis: $(g+h\,x)^m == \partial_x \frac{(g+h\,x)^{m+1}}{h\,(m+1)}$

Basis: If $p+q==0$, then $\partial_x \text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s == \frac{p\,r\,s\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,p+q==0\,\wedge\,s\in\mathbb{Z}^+\,\wedge\,m\neq -1$, then

$$\int (g+h\,x)^m\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx\,\longrightarrow$$

$$\frac{(g+h\,x)^{m+1}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{h\,(m+1)}-\frac{p\,r\,s\,(b\,c-a\,d)}{h\,(m+1)}\int\frac{(g+h\,x)^{m+1}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(h*(m+1)) -
  p*r*s*(b*c-a*d)/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && IGtQ[s,0] && NeQ[m,-1]
```

5: $\int \frac{1}{(g+h\,x)^2\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,p+q==0\,\wedge\,b\,g-a\,h==0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,p+q==0\,\wedge\,b\,g-a\,h==0$, then

$$\int \frac{1}{(g+h\,x)^2\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}\,dx\,\longrightarrow$$

$$\frac{b\,(c+d\,x)\,\left(e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right)^{\frac{1}{p\,r}}}{h\,p\,r\,(b\,c-a\,d)\,(g+h\,x)}\,\text{ExpIntegralEi}\left[-\frac{1}{p\,r}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]\right]$$

Program code:

```
Int[1/((g_.+h_.*x_)^2*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]),x_Symbol] :=
  b*(c+d*x)*(e*(f*(a+b*x)^p*(c+d*x)^q)^r)^(1/(p*r))/(h*p*r*(b*c-a*d)*(g+h*x))*
    ExpIntegralEi[-1/(p*r)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[b*g-a*h,0]
```

2. $\int (g+h\,x)^m \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{g+h\,x} = \partial_x \frac{\operatorname{Log}[g+h\,x]}{h}$

Basis: $\partial_x \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right] = \frac{b\,p\,r}{a+b\,x} + \frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{g+h\,x}\,dx \longrightarrow$$

$$\frac{\operatorname{Log}[g+h\,x]\,\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{h} - \frac{b\,p\,r}{h}\int \frac{\operatorname{Log}[g+h\,x]}{a+b\,x}\,dx - \frac{d\,q\,r}{h}\int \frac{\operatorname{Log}[g+h\,x]}{c+d\,x}\,dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/(g_.+h_.*x_),x_Symbol] :=
  Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/h -
  b*p*r/h*Int[Log[g+h*x]/(a+b*x),x] -
  d*q*r/h*Int[Log[g+h*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0]
```

2: $\int (g+h\,x)^m \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right] dx$ when $b\,c-a\,d\neq 0\,\land\,m\neq -1$

Derivation: Integration by parts

Basis: $(g+h\,x)^m == \partial_x \frac{(g+h\,x)^{m+1}}{h\,(m+1)}$

Basis: $\partial_x \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right] == \frac{b\,p\,r}{a+b\,x}+\frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\,\land\,m\neq -1$, then

$$\int (g+h\,x)^m \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right] dx \longrightarrow$$

$$\frac{(g+h\,x)^{m+1}\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{h\,(m+1)}-\frac{b\,p\,r}{h\,(m+1)}\int\frac{(g+h\,x)^{m+1}}{a+b\,x}dx-\frac{d\,q\,r}{h\,(m+1)}\int\frac{(g+h\,x)^{m+1}}{c+d\,x}dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  (g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(h*(m+1)) -
  b*p*r/(h*(m+1))*Int[(g+h*x)^(m+1)/(a+b*x),x] -
  d*q*r/(h*(m+1))*Int[(g+h*x)^(m+1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,p,q,r},x] && NeQ[b*c-a*d,0] && NeQ[m,-1]
```

3. $\int \frac{\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^2}{g+hx}\,dx$ when $bc-ad\neq 0$

1: $\int \frac{\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^2}{g+hx}\,dx$ when $bc-ad\neq 0\land bg-ah==0$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left(\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]-\text{Log}\left[(a+bx)^{pr}\right]-\text{Log}\left[(c+dx)^{qr}\right]\right)==0$

Rule: If $bc-ad\neq 0\land bg-ah==0$, then

$$\int \frac{\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^2}{g+hx}\,dx\longrightarrow$$

$$\int \frac{\left(\text{Log}\left[(a+bx)^{pr}\right]+\text{Log}\left[(c+dx)^{qr}\right]\right)^2}{g+hx}\,dx+\left(\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]-\text{Log}\left[(a+bx)^{pr}\right]-\text{Log}\left[(c+dx)^{qr}\right]\right)\cdot$$
$$\left(2\int \frac{\text{Log}\left[(c+dx)^{qr}\right]}{g+hx}\,dx+\int \frac{1}{g+hx}\left(\text{Log}\left[(a+bx)^{pr}\right]-\text{Log}\left[(c+dx)^{qr}\right]+\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]\right)dx\right)$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^2/(g_.+h_.*x_),x_Symbol] :=
  Int[(Log[(a+b*x)^(p*r)]+Log[(c+d*x)^(q*r)])^2/(g+h*x),x] +
  (Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]-Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)])*
    (2*Int[Log[(c+d*x)^(q*r)]/(g+h*x),x] +
     Int[(Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)]+Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r])/(g+h*x),x]) /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[b*g-a*h,0]
```

1: $\int \frac{\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^2}{g+h x} \, dx$ when $b c-a d \neq 0 \land b g-a h \neq 0 \land d g-c h \neq 0$?? ??

Derivation: Piecewise constant extraction

Basis: $\partial_x \left(\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]-\text{Log}\left[(a+b x)^{p r}\right]-\text{Log}\left[(c+d x)^{q r}\right]\right) == 0$

Rule: If $b c-a d \neq 0 \land b g-a h == 0$, then

$$\int \frac{\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^2}{g+h x} \, dx \to$$

$$\int \frac{\left(\text{Log}\left[(a+b x)^{p r}\right]+\text{Log}\left[(c+d x)^{q r}\right]\right)^2}{g+h x} \, dx +$$
$$\left(\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]-\text{Log}\left[(a+b x)^{p r}\right]-\text{Log}\left[(c+d x)^{q r}\right]\right)$$
$$\int \frac{\text{Log}\left[(a+b x)^{p r}\right]+\text{Log}\left[(c+d x)^{q r}\right]+\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]}{g+h x} \, dx$$

Program code:

```
(* Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^2/(g_.+h_.*x_),x_Symbol] :=
  Int[(Log[(a+b*x)^(p*r)]+Log[(c+d*x)^(q*r)])^2/(g+h*x),x] +
  (Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]-Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)])*
    Int[(Log[(a+b*x)^(p*r)]+Log[(c+d*x)^(q*r)]+Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r])/(g+h*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[b*g-a*h,0] *)
```

2: $\int \frac{\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^2}{g+h x} \, dx$ when $b c-a d \neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{g+h x} == \partial_x \frac{\text{Log}[g+h x]}{h}$

Basis:

$$\partial_x \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^2 == \frac{2 b p r \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]}{a+b x}+\frac{2 d q r \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]}{c+d x}$$

Rule: If $b c - a d \neq 0$, then

$$\int \frac{\operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^2}{g+h x} \, dx \longrightarrow$$

$$\frac{\operatorname{Log}[g+h x] \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^2}{h}-$$

$$\frac{2 b p r}{h} \int \frac{\operatorname{Log}[g+h x] \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]}{a+b x} \, dx-\frac{2 d q r}{h} \int \frac{\operatorname{Log}[g+h x] \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]}{c+d x} \, dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^2/(g_.+h_.*x_),x_Symbol] :=
  Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^2/h -
  2*b*p*r/h*Int[Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(a+b*x),x] -
  2*d*q*r/h*Int[Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0]
```

4: $\int (g+h x)^m \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s \, dx$ when $b c - a d \neq 0 \land s \in \mathbb{Z}^+ \land m \neq -1$

Derivation: Integration by parts

Basis: $(g+h x)^m == \partial_x \frac{(g+h x)^{m+1}}{h (m+1)}$

Basis: $\partial_x \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s == \frac{b p r s}{a+b x} \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}+\frac{d q r s}{c+d x} \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c - a d \neq 0 \land s \in \mathbb{Z}^+ \land m \neq -1$, then

$$\int (g+h x)^m \operatorname{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s \, dx \longrightarrow$$

$$\frac{(g+h\,x)^{m+1}\,\mathrm{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{h\,(m+1)}-$$

$$\frac{b\,p\,r\,s}{h\,(m+1)}\int\frac{(g+h\,x)^{m+1}\,\mathrm{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{a+b\,x}\,dx-\frac{d\,q\,r\,s}{h\,(m+1)}\int\frac{(g+h\,x)^{m+1}\,\mathrm{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{c+d\,x}\,dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_,x_Symbol] :=
  (g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(h*(m+1)) -
  b*p*r*s/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(a+b*x),x] -
  d*q*r*s/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && NeQ[m,-1]
```

4. $\int \frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^m\,\mathrm{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^u}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^m\,\mathrm{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,h\,j-g\,k==0\,\wedge\,m\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: If $h\,j-g\,k==0$, then $\frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^m}{j+k\,x}==\partial_x\frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^{m+1}}{k\,n\,t\,(m+1)}$

Basis: $\partial_x\,\mathrm{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]==\frac{b\,p\,r}{a+b\,x}+\frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,h\,j-g\,k==0\,\wedge\,m\in\mathbb{Z}^+$, then

$$\int \frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^m\,\mathrm{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]}{j+k\,x}\,dx\,\longrightarrow$$

$$\frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^{m+1}\,\mathrm{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]}{k\,n\,t\,(m+1)}-$$
$$\frac{b\,p\,r}{k\,n\,t\,(m+1)}\int\frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^{m+1}}{a+b\,x}\,dx-\frac{d\,q\,r}{k\,n\,t\,(m+1)}\int\frac{(s+t\,\mathrm{Log}[i\,(g+h\,x)^n])^{m+1}}{c+d\,x}\,dx$$

Program code:

```
Int[(s_.+t_.*Log[i_.*(g_.+h_.*x_)^n_.])^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/(j_.+k_.*x_),x_Symbol] :=
  (s+t*Log[i*(g+h*x)^n])^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(k*n*t*(m+1)) -
  b*p*r/(k*n*t*(m+1))*Int[(s+t*Log[i*(g+h*x)^n])^(m+1)/(a+b*x),x] -
  d*q*r/(k*n*t*(m+1))*Int[(s+t*Log[i*(g+h*x)^n])^(m+1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,s,t,m,n,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[h*j-g*k,0] && IGtQ[m,0]
```

2: $\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left(\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]-\text{Log}[(a+b\,x)^{p\,r}]-\text{Log}[(c+d\,x)^{q\,r}]\right)=0$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]}{j+k\,x}\,dx \to$$

$$\left(\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]-\text{Log}[(a+b\,x)^{p\,r}]-\text{Log}[(c+d\,x)^{q\,r}]\right)\int \frac{s+t\,\text{Log}[i\,(g+h\,x)^n]}{j+k\,x}\,dx+$$
$$\int \frac{\text{Log}[(a+b\,x)^{p\,r}]\,(s+t\,\text{Log}[i\,(g+h\,x)^n])}{j+k\,x}\,dx+\int \frac{\text{Log}[(c+d\,x)^{q\,r}]\,(s+t\,\text{Log}[i\,(g+h\,x)^n])}{j+k\,x}\,dx$$

Program code:

```
Int[(s_.+t_.*Log[i_.*(g_.+h_.*x_)^n_.])*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/(j_.+k_.*x_),x_Symbol] :=
  (Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]-Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)])*Int[(s+t*Log[i*(g+h*x)^n])/(j+k*x),x] +
  Int[(Log[(a+b*x)^(p*r)]*(s+t*Log[i*(g+h*x)^n]))/(j+k*x),x] +
  Int[(Log[(c+d*x)^(q*r)]*(s+t*Log[i*(g+h*x)^n]))/(j+k*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,s,t,n,p,q,r},x] && NeQ[b*c-a*d,0]
```

U: $\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^u}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^u}{j+k\,x}\,dx \longrightarrow \int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^u}{j+k\,x}\,dx$$

Program code:

```
Int[(s_.+t_.*Log[i_.*(g_.+h_.*x_)^n_.])^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^u_./(j_.+k_.*x_),x_Symbol] :=
  Unintegrable[(s+t*Log[i*(g+h*x)^n])^m*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^u/(j+k*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,s,t,m,n,p,q,r,u},x] && NeQ[b*c-a*d,0]
```

5. $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(g+h\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\,\wedge\,p+q==0$

1: $\int \frac{\text{Log}\left[e\,\frac{c+d\,x}{a+b\,x}\right]}{(a+b\,x)\,(g+h\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,g\,(b-d\,e)-h\,(a-c\,e)==0$

Derivation: Integration by substitution

Basis: If $g\,(b-d\,e)-h\,(a-c\,e)==0$, then $\frac{\text{Log}\left[e\,\frac{c+d\,x}{a+b\,x}\right]}{(a+b\,x)\,(g+h\,x)}==-\frac{b-d\,e}{h\,(b\,c-a\,d)}\,\text{Subst}\left[\frac{\text{Log}[e\,x]}{1-e\,x},\,x,\,\frac{c+d\,x}{a+b\,x}\right]\,\partial_x\,\frac{c+d\,x}{a+b\,x}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,g\,(b-d\,e)-h\,(a-c\,e)==0$, then

$$\int \frac{\text{Log}\left[e\,\frac{c+d\,x}{a+b\,x}\right]}{(a+b\,x)\,(g+h\,x)}\,dx\;\longrightarrow\;-\frac{b-d\,e}{h\,(b\,c-a\,d)}\,\text{Subst}\left[\int\frac{\text{Log}[e\,x]}{1-e\,x}\,dx,\,x,\,\frac{c+d\,x}{a+b\,x}\right]$$

Program code:

```
Int[u_*Log[e_.*(c_.+d_.*x_)/(a_.+b_.*x_)],x_Symbol] :=
  With[{g=Coeff[Simplify[1/(u*(a+b*x))],x,0],h=Coeff[Simplify[1/(u*(a+b*x))],x,1]},
  -(b-d*e)/(h*(b*c-a*d))*Subst[Int[Log[e*x]/(1-e*x),x],x,(c+d*x)/(a+b*x)] /;
 EqQ[g*(b-d*e)-h*(a-c*e),0]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*c-a*d,0] && LinearQ[Simplify[1/(u*(a+b*x))],x]
```

2: $\int \frac{\operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s}{(a+bx)(g+hx)}\,dx$ when $bc-ad\neq 0\land s\in\mathbb{Z}^+\land p+q=0\land bg-ah\neq 0\land dg-ch\neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{(a+bx)(g+hx)} = -\frac{1}{bg-ah}\partial_x \operatorname{Log}\left[-\frac{(bc-ad)(g+hx)}{(dg-ch)(a+bx)}\right]$

Basis: If $p+q=0$, then $\partial_x \operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s = \frac{prs(bc-ad)}{(a+bx)(c+dx)}\operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^{s-1}$

Rule: If $bc-ad\neq 0\land s\in\mathbb{Z}^+\land p+q=0\land bg-ah\neq 0\land dg-ch\neq 0$, then

$$\int \frac{\operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s}{(a+bx)(g+hx)}\,dx \to$$

$$-\frac{\operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s}{bg-ah}\operatorname{Log}\left[-\frac{(bc-ad)(g+hx)}{(dg-ch)(a+bx)}\right]+$$
$$\frac{prs(bc-ad)}{bg-ah}\int \frac{\operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^{s-1}}{(a+bx)(c+dx)}\operatorname{Log}\left[-\frac{(bc-ad)(g+hx)}{(dg-ch)(a+bx)}\right]dx$$

Program code:

```
Int[u_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Coeff[Simplify[1/(u*(a+b*x))],x,0],h=Coeff[Simplify[1/(u*(a+b*x))],x,1]},
  -Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*g-a*h)*Log[-(b*c-a*d)*(g+h*x)/((d*g-c*h)*(a+b*x))] +
  p*r*s*(b*c-a*d)/(b*g-a*h)*
    Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x))*Log[-(b*c-a*d)*(g+h*x)/((d*g-c*h)*(a+b*x))],x] /;
 NeQ[b*g-a*h,0] && NeQ[d*g-c*h,0]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0] && LinearQ[Simplify[1/(u*(a+b*x))],x]
```

6. $\int \frac{u\operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s}{(a+bx)(c+dx)}\,dx$ when $bc-ad\neq 0\land p+q=0$

1. $\int \frac{\text{Log}\left[e\left(f(a+b x)^{p}(c+d x)^{q}\right)^{r}\right]^{s}}{(a+b x)(c+d x)} \, dx$ when $b c-a d \neq 0 \wedge p+q==0$

1: $\int \frac{1}{(a+b x)(c+d x) \text{Log}\left[e\left(f(a+b x)^{p}(c+d x)^{q}\right)^{r}\right]} \, dx$ when $b c-a d \neq 0 \wedge p+q==0$

Rule: If $b c-a d \neq 0 \wedge p+q==0$, then

$$\int \frac{1}{(a+b x)(c+d x) \text{Log}\left[e\left(f(a+b x)^{p}(c+d x)^{q}\right)^{r}\right]} \, dx \longrightarrow \frac{\text{Log}\left[\text{Log}\left[e\left(f(a+b x)^{p}(c+d x)^{q}\right)^{r}\right]\right]}{p r(b c-a d)}$$

Program code:

```
Int[u_/Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  With[{h=Simplify[u*(a+b*x)*(c+d*x)]},
  h*Log[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]]/(p*r*(b*c-a*d)) /;
 FreeQ[h,x]] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0]
```

2: $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,p+q==0\,\land\,s\neq -1$

Rule: If $b\,c-a\,d\neq 0\,\land\,p+q==0\,\land\,s\neq -1$, then

$$\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx \longrightarrow \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s+1}}{p\,r\,(s+1)\,(b\,c-a\,d)}$$

Program code:

```
Int[u_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{h=Simplify[u*(a+b*x)*(c+d*x)]},
  h*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(p*r*(s+1)*(b*c-a*d)) /;
 FreeQ[h,x]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && NeQ[s,-1]
```

2: $\int \frac{\text{Log}\left[1+g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\,\land\,p+q==0$

Derivation: Integration by parts

Basis: $\frac{\text{Log}\left[1+g\,\frac{a+b\,x}{c+d\,x}\right]}{(a+b\,x)\,(c+d\,x)} == -\partial_x\,\frac{\text{PolyLog}\left[2,-g\,\frac{a+b\,x}{c+d\,x}\right]}{b\,c-a\,d}$

Basis: If $p+q==0$, then $\partial_x\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s == \frac{p\,r\,s\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}$

Rule: If $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\,\land\,p+q==0$, then

$$\int \frac{\text{Log}\left[1+g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx \longrightarrow$$

$$-\frac{\text{PolyLog}\left[2,-g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{b\,c-a\,d}+p\,r\,s\int\frac{\text{PolyLog}\left[2,-g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[u_*Log[v_]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Simplify[(v-1)*(c+d*x)/(a+b*x)],h=Simplify[u*(a+b*x)*(c+d*x)]},
  -h*PolyLog[2,1-v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*c-a*d) +
  h*p*r*s*Int[PolyLog[2,1-v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
 FreeQ[{g,h},x]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0]
```

3: $\int \frac{\text{Log}\left[i\left(j\left(g+h\,x\right)^t\right)^u\right]\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]^s}{\left(a+b\,x\right)\left(c+d\,x\right)}\,dx$ when $b\,c-a\,d\neq 0\wedge p+q==0\wedge s\neq -1$

Derivation: Integration by parts

Basis: If $p+q==0$, then $\frac{\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]^s}{\left(a+b\,x\right)\left(c+d\,x\right)}==\partial_x\frac{\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]^{s+1}}{p\,r\,(s+1)\,(b\,c-a\,d)}$

Basis: $\partial_x\text{Log}\left[i\left(j\left(g+h\,x\right)^t\right)^u\right]==\frac{h\,t\,u}{g+h\,x}$

Rule: If $b\,c-a\,d\neq 0\wedge p+q==0\wedge s\neq -1$, then

$$\int \frac{\text{Log}\left[i\left(j\left(g+h\,x\right)^t\right)^u\right]\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]^s}{\left(a+b\,x\right)\left(c+d\,x\right)}\,dx\to$$

$$\frac{\text{Log}\left[i\left(j\left(g+h\,x\right)^t\right)^u\right]\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]^{s+1}}{p\,r\,(s+1)\,\left(b\,c-a\,d\right)}-\frac{h\,t\,u}{p\,r\,(s+1)\,\left(b\,c-a\,d\right)}\int\frac{\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]^{s+1}}{g+h\,x}\,dx$$

Program code:

```
Int[v_*Log[i_.*(j_.*(g_.+h_.*x_)^t_.)^u_.]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{k=Simplify[v*(a+b*x)*(c+d*x)]},
  k*Log[i*(j*(g+h*x)^t)^u]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(p*r*(s+1)*(b*c-a*d)) -
  k*h*t*u/(p*r*(s+1)*(b*c-a*d))*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(g+h*x),x] /;
 FreeQ[k,x]] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,p,q,r,s,t,u},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && NeQ[s,-1]
```

4: $\int \frac{\text{PolyLog}\left[n,\,g\,\frac{a+b\,x}{c+d\,x}\right]\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]^s}{\left(a+b\,x\right)\left(c+d\,x\right)}\,dx$ when $b\,c-a\,d\neq 0\wedge s\in\mathbb{Z}^+\wedge p+q==0$

Derivation: Integration by parts

Basis: $\frac{\text{PolyLog}\left[n,g\frac{a+b x}{c+d x}\right]}{(a+b x)(c+d x)} = \partial_x \frac{\text{PolyLog}\left[n+1,g\frac{a+b x}{c+d x}\right]}{b c-a d}$

Basis: If $p+q=0$, then $\partial_x \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s = \frac{p r s (b c-a d)}{(a+b x)(c+d x)} \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d \neq 0 \land s \in \mathbb{Z}^+ \land p+q=0$, then

$$\int \frac{\text{PolyLog}\left[n, g\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s}{(a+b x)(c+d x)} \, dx \longrightarrow$$

$$\frac{\text{PolyLog}\left[n+1, g\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s}{b c-a d} - p r s \int \frac{\text{PolyLog}\left[n+1, g\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{(a+b x)(c+d x)} \, dx$$

Program code:

```
Int[u_*PolyLog[n_,v_]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Simplify[v*(c+d*x)/(a+b*x)],h=Simplify[u*(a+b*x)*(c+d*x)]},
  h*PolyLog[n+1,v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*c-a*d) -
  h*p*r*s*Int[PolyLog[n+1,v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
 FreeQ[{g,h},x]] /;
FreeQ[{a,b,c,d,e,f,n,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0]
```

7. $\int (a+b x)^m (c+d x)^n \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0 \land p+q=0 \land m+n+2=0 \land m \neq -1$

1: $\int (a+b x)^m (c+d x)^n \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0 \land p+q=0 \land m+n+2=0 \land m \neq -1 \land s \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: If $m+n+2=0$, then $(a+b x)^m (c+d x)^n = \partial_x \frac{(a+b x)^{m+1} (c+d x)^{n+1}}{(m+1)(b c-a d)}$

Basis: If $p+q=0$, then $\partial_x \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s = \frac{p r s (b c-a d)}{(a+b x)(c+d x)} \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d \neq 0 \land p+q=0 \land m+n+2=0 \land m \neq -1 \land s \in \mathbb{Z}^+$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx\;\longrightarrow$$

$$\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(m+1)\,(b\,c-a\,d)}-\frac{p\,r\,s\,(b\,c-a\,d)}{(m+1)\,(b\,c-a\,d)}\int (a+b\,x)^m\,(c+d\,x)^n\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/((m+1)*(b*c-a*d)) -
  p*r*s*(b*c-a*d)/((m+1)*(b*c-a*d))*Int[(a+b*x)^m*(c+d*x)^n*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[m+n+2,0] && NeQ[m,-1] && IGtQ[s,0]
```

2: $\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,p+q==0\,\land\,m+n+2==0\,\land\,m\neq -1$

Rule: If $b\,c-a\,d\neq 0\,\land\,p+q==0\,\land\,m+n+2==0\,\land\,m\neq -1$, then

$$\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}\,dx\;\longrightarrow$$

$$\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{p\,r\,(b\,c-a\,d)\,\left(e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right)^{\frac{m+1}{p\,r}}}\,\text{ExpIntegralEi}\left[\frac{m+1}{p\,r}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]\right]$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_./Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/(p*r*(b*c-a*d)*(e*(f*(a+b*x)^p*(c+d*x)^q)^r)^((m+1)/(p*r)))*
    ExpIntegralEi[(m+1)/(p*r)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[m+n+2,0] && NeQ[m,-1]
```

8: $\int \frac{\left(a+b\,\mathrm{Log}\left[c\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\right)^n}{A+B\,x+C\,x^2}\,dx$ when $C\,d\,f-A\,e\,g==0\,\land\,B\,e\,g-C\,(e\,f+d\,g)==0\,\land\,n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x]==2\,(e\,f-d\,g)\,\mathrm{Subst}\left[\frac{x}{(e-g\,x^2)^2}\,F\left[-\frac{d-f\,x^2}{e-g\,x^2}\right],\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\partial_x\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}$

Basis: If $C\,d\,f-A\,e\,g==0\,\land\,B\,e\,g-C\,(e\,f+d\,g)==0$, then

$\frac{1}{A+B\,x+C\,x^2}==\frac{2\,e\,g}{C\,(e\,f-d\,g)}\,\mathrm{Subst}\left[\frac{1}{x},\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\partial_x\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}$

Rule: If $C\,d\,f-A\,e\,g==0\,\land\,B\,e\,g-C\,(e\,f+d\,g)==0\,\land\,n\in\mathbb{Z}^+$, then

$$\int \frac{\left(a+b\,\mathrm{Log}\left[c\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\right)^n}{A+B\,x+C\,x^2}\,dx\,\longrightarrow\,\frac{2\,e\,g}{C\,(e\,f-d\,g)}\,\mathrm{Subst}\left[\int\frac{(a+b\,\mathrm{Log}[c\,x])^n}{x}\,dx,\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]])^n_./(A_.+B_.*x_+C_.*x_^2),x_Symbol] :=
  2*e*g/(C*(e*f-d*g))*Subst[Int[(a+b*Log[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,B,C,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[B*e*g-C*(e*f+d*g),0]
```

```
Int[(a_.+b_.*Log[c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]])^n_./(A_.+C_.*x_^2),x_Symbol] :=
  g/(C*f)*Subst[Int[(a+b*Log[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,C,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[e*f+d*g,0]
```

9. $\int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$

1: $\int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx$ when $b c - a d \neq 0$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $u A == u B + u C - (B + C - A) u$

Basis: $\partial_x \left(p r \operatorname{Log}[a+b x] + q r \operatorname{Log}[c+d x] - \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]\right) == 0$

Rule: If $b c - a d \neq 0$, then

$$\int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx \to$$

$$p r \int RF_x \operatorname{Log}[a+b x] dx + q r \int RF_x \operatorname{Log}[c+d x] dx - \left(p r \operatorname{Log}[a+b x] + q r \operatorname{Log}[c+d x] - \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]\right) \int RF_x dx$$

Program code:

```
Int[RFx_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  p*r*Int[RFx*Log[a+b*x],x] +
  q*r*Int[RFx*Log[c+d*x],x] -
  (p*r*Log[a+b*x]+q*r*Log[c+d*x] - Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r])*Int[RFx,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && RationalFunctionQ[RFx,x] && NeQ[b*c-a*d,0] &&
  Not[MatchQ[RFx,u_.*(a+b*x)^m_.*(c+d*x)^n_. /; IntegersQ[m,n]]]
```

x: $\int RF_x \, \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx$ when $b c-a d \neq 0$

Derivation: Integration by parts

Basis: $\partial_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] = \frac{b p r}{a+b x}+\frac{d q r}{c+d x}$

Rule: If $b c-a d \neq 0$, let $u \to \int RF_x \, dx$, then

$$\int RF_x \, \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx \longrightarrow u \, \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]-b p r \int \frac{u}{a+b x} dx-d q r \int \frac{u}{c+d x} dx$$

Program code:

```
(* Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  With[{u=IntHide[RFx,x]},
  u*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r] - b*p*r*Int[u/(a+b*x),x] - d*q*r*Int[u/(c+d*x),x] /;
 NonsumQ[u]] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && RationalFunctionQ[RFx,x] && NeQ[b*c-a*d,0] *)
```

2: $\int RF_x \,\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s dx$ when $s\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $s\in\mathbb{Z}^+$, then

$$\int RF_x \,\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s dx \to \int \text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s \,\text{ExpandIntegrand}[RF_x,\,x]\,dx$$

Program code:

```
Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && RationalFunctionQ[RFx,x] && IGtQ[s,0]
```

U: $\int RF_x \,\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s dx$

Rule:

$$\int RF_x \,\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s dx \to \int RF_x \,\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s dx$$

Program code:

```
Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  Unintegrable[RFx*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && RationalFunctionQ[RFx,x]
```

N: $\int u \,\mathrm{Log}\left[e \left(f\, v^p\, w^q\right)^r\right]^s dx$ when $v == a + b\, x \land w == c + d\, x$

Derivation: Algebraic normalization

Rule: If $v == a + b\, x \land w == c + d\, x$, then

$$\int u \,\mathrm{Log}\left[e \left(f\, v^p\, w^q\right)^r\right]^s dx \longrightarrow \int u \,\mathrm{Log}\left[e \left(f\, (a + b\, x)^p\, (c + d\, x)^q\right)^r\right]^s dx$$

Program code:

```
Int[u_.*Log[e_.*(f_.*v_^p_.*w_^q_.)^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(f*ExpandToSum[v,x]^p*ExpandToSum[w,x]^q)^r]^s,x] /;
FreeQ[{e,f,p,q,r,s},x] && LinearQ[{v,w},x] && Not[LinearMatchQ[{v,w},x]] && AlgebraicFunctionQ[u,x]
```

```
Int[u_.*Log[e_.*(f_.*(g_+v_./w_))^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(f*ExpandToSum[v+g*w,x]/ExpandToSum[w,x])^r]^s,x] /;
FreeQ[{e,f,g,r,s},x] && LinearQ[w,x] && (FreeQ[v,x] || LinearQ[v,x]) && AlgebraicFunctionQ[u,x]
```

x: $\int \frac{\text{Log}\left[i\left(j\left(g+h\,x\right)^s\right)^t\right]\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]}{m+n\,x}\,dx$

Derivation: Integration by substitution

Basis: $F[x] = \frac{1}{n}\,\text{Subst}\left[F\left[\frac{x-m}{n}\right],\,x,\,m+n\,x\right]\,\partial_x\,(m+n\,x)$

Rule:

$$\int \frac{\text{Log}\left[i\left(j\left(g+h\,x\right)^s\right)^t\right]\text{Log}\left[e\left(f\left(a+b\,x\right)^p\left(c+d\,x\right)^q\right)^r\right]}{m+n\,x}\,dx \longrightarrow$$

$$\frac{1}{n}\,\text{Subst}\left[\int \frac{1}{x}\text{Log}\left[i\left(j\left(-\frac{h\,m-g\,n}{n}+\frac{h\,x}{n}\right)^s\right)^t\right]\text{Log}\left[e\left(f\left(-\frac{b\,m-a\,n}{n}+\frac{b\,x}{n}\right)^p\left(-\frac{d\,m-c\,n}{n}+\frac{d\,x}{n}\right)^q\right)^r\right]dx,\,x,\,m+n\,x\right]$$

Program code:

```
(* Int[Log[g_.*(h_.*(a_.+b_.*x_)^p_.)^q_.]*Log[i_.*(j_.*(c_.+d_.*x_)^r_.)^s_.]/(e_+f_.*x_),x_Symbol] :=
  1/f*Subst[Int[Log[g*(h*Simp[-(b*e-a*f)/f+b*x/f,x]^p)^q]*Log[i*(j*Simp[-(d*e-c*f)/f+d*x/f,x]^r)^s]/x,x],x,e+f*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,p,q,r,s},x] *)
```