

Rules for integrands of the form $(a + b \sin[ex + fx])^m (A + B \sin[ex + fx] + C \sin[ex + fx]^2)$

1: $\int (b \sin[ex + fx])^m (B \sin[ex + fx] + C \sin[ex + fx]^2) dx$

Derivation: Algebraic simplification

Rule:

$$\int (b \sin[ex + fx])^m (B \sin[ex + fx] + C \sin[ex + fx]^2) dx \rightarrow \frac{1}{b} \int (b \sin[ex + fx])^{m+1} (B + C \sin[ex + fx]) dx$$

Program code:

```
Int[(b_.sin[e_.+f_.x_])^m_.*(B_.sin[e_.+f_.x_]+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  1/b*Int[(b*sin[e+f*x])^(m+1)*(B+C*sin[e+f*x]),x] /;
FreeQ[{b,e,f,B,C,m},x]
```

2. $\int (b \sin[ex + fx])^m (A + C \sin[ex + fx]^2) dx$

1: $\int (b \sin[ex + fx])^m (A + C \sin[ex + fx]^2) dx$ when $A(m+2) + C(m+1) = 0$

Derivation: Nondegenerate sine recurrence 1a with $n \rightarrow 0$, $p \rightarrow 0$

Rule: If $A(m+2) + C(m+1) = 0$, then

$$\int (b \sin[ex + fx])^m (A + C \sin[ex + fx]^2) dx \rightarrow \frac{A \cos[ex + fx] (b \sin[ex + fx])^{m+1}}{b f (m+1)}$$

Program code:

```
Int[(b_.sin[e_.+f_.x_])^m_.*(A+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  A*cos[e+f*x]*(b*sin[e+f*x])^(m+1)/(b*f*(m+1)) /;
FreeQ[{b,e,f,A,C,m},x] && EqQ[A*(m+2)+C*(m+1),0]
```

2: $\int (b \sin[e + f x])^m (A + C \sin[e + f x]^2) dx$ when $m < -1$

Derivation: Nondegenerate sine recurrence 1a with $n \rightarrow 0$, $p \rightarrow 0$

Rule: If $m < -1$, then

$$\int (b \sin[e + f x])^m (A + C \sin[e + f x]^2) dx \rightarrow \frac{A \cos[e + f x] (b \sin[e + f x])^{m+1}}{b f (m+1)} + \frac{A (m+2) + C (m+1)}{b^2 (m+1)} \int (b \sin[e + f x])^{m+2} dx$$

Program code:

```
Int[(b_.sin[e_.+f_.*x_])^m_*(A_+C_.sin[e_.+f_.*x_]^2),x_Symbol] :=
  A*cos[e+f*x]*(b*sin[e+f*x])^(m+1)/(b*f*(m+1)) + (A*(m+2)+C*(m+1))/(b^2*(m+1))*Int[(b*sin[e+f*x])^(m+2),x] /;
FreeQ[{b,e,f,A,C},x] && LtQ[m,-1]
```

$$3. \int (b \sin[e + f x])^m (A + C \sin[e + f x]^2) dx \text{ when } m \neq -1$$

$$1: \int \sin[e + f x]^m (A + C \sin[e + f x]^2) dx \text{ when } \frac{m+1}{2} \in \mathbb{Z}^+$$

Derivation: Algebraic expansion and integration by substitution

$$\text{Basis: } \sin[z]^2 = 1 - \cos[z]^2$$

$$\text{Basis: If } \frac{m+1}{2} \in \mathbb{Z}, \text{ then } \sin[e + f x]^m = -\frac{1}{f} \text{Subst}\left[\left(1 - x^2\right)^{\frac{m-1}{2}}, x, \cos[e + f x]\right] \partial_x \cos[e + f x]$$

Rule: If $\frac{m+1}{2} \in \mathbb{Z}^+$, then

$$\begin{aligned} \int \sin[e + f x]^m (A + C \sin[e + f x]^2) dx &\rightarrow \int \sin[e + f x]^m (A + C - C \cos[e + f x]^2) dx \\ &\rightarrow -\frac{1}{f} \text{Subst}\left[\int (1 - x^2)^{\frac{m-1}{2}} (A + C - C x^2) dx, x, \cos[e + f x]\right] \end{aligned}$$

Program code:

```
Int[sin[e_+f_*x_]^m_.*(A_+C_*sin[e_+f_*x_]^2),x_Symbol] :=
  -1/f*Subst[Int[(1-x^2)^( (m-1)/2)*(A+C-C*x^2),x],x,Cos[e+f*x]] /;
FreeQ[{e,f,A,C},x] && IGtQ[(m+1)/2,0]
```

2: $\int (b \sin[e + f x])^m (A + C \sin[e + f x]^2) dx$ when $m \neq -1$

Derivation: Nondegenerate sine recurrence 1b with $m \rightarrow 0$, $p \rightarrow 0$

Rule: If $m \neq -1$, then

$$\int (b \sin[e + f x])^m (A + C \sin[e + f x]^2) dx \rightarrow$$

$$- \frac{C \cos[e + f x] (b \sin[e + f x])^{m+1}}{b f (m+2)} + \frac{A (m+2) + C (m+1)}{m+2} \int (b \sin[e + f x])^m dx$$

Program code:

```
Int[(b_.sin[e_.+f_.x_])^m_.*(A_+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  -C*cos[e+f*x]*(b*sin[e+f*x])^(m+1)/(b*f*(m+2)) + (A*(m+2)+C*(m+1))/(m+2)*Int[(b*sin[e+f*x])^m,x] /;
FreeQ[{b,e,f,A,C,m},x] && Not[LtQ[m,-1]]
```

3: $\int (a + b \sin[e + f x])^m (A + B \sin[e + f x] + C \sin[e + f x]^2) dx$ when $A b^2 - a b B + a^2 C = 0$

Derivation: Algebraic simplification

Basis: If $A b^2 - a b B + a^2 C = 0$, then $A + B z + C z^2 = \frac{1}{b^2} (a + b z) (b B - a C + b C z)$

Rule: If $a^2 - b^2 \neq 0 \wedge A b^2 - a b B + a^2 C = 0$, then

$$\int (a + b \sin[e + f x])^m (A + B \sin[e + f x] + C \sin[e + f x]^2) dx \rightarrow \frac{1}{b^2} \int (a + b \sin[e + f x])^{m+1} (b B - a C + b C \sin[e + f x]) dx$$

Program code:

```
Int[(a_+b_.sin[e_+f_.x_])^m_.*(A_+B_.sin[e_+f_.x_]+C_.sin[e_+f_.x_]^2),x_Symbol] :=
  1/b^2*Int[(a+b*sin[e+f*x])^(m+1)*Simp[b*B-a*C+b*C*sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && EqQ[A*b^2-a*b*B+a^2*C,0]
```

```
Int[(a_+b_.sin[e_+f_.x_])^m_.*(A_+C_.sin[e_+f_.x_]^2),x_Symbol] :=
  C/b^2*Int[(a+b*sin[e+f*x])^(m+1)*Simp[-a+b*sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A*b^2+a^2*C,0]
```

4: $\int (a+b \sin[e+f x])^m (A+B \sin[e+f x]+C \sin[e+f x]^2) dx$ when $A-B+C=0 \wedge 2m \notin \mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $A-B+C=0$, then $A+Bz+Cz^2=(A-C)(1+z)+C(1+z)^2$

Rule: If $A-B+C=0 \wedge 2m \notin \mathbb{Z}$, then

$$\int (a+b \sin[e+f x])^m (A+B \sin[e+f x]+C \sin[e+f x]^2) dx \rightarrow (A-C) \int (a+b \sin[e+f x])^m (1+\sin[e+f x]) dx + C \int (a+b \sin[e+f x])^m (1+\sin[e+f x])^2 dx$$

Program code:

```
Int[(a+b_.sin[e_.+f_.x_])^m_.*(A_.+B_.sin[e_.+f_.x_]+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  (A-C)*Int[(a+b*sin[e+f*x])^m*(1+Sin[e+f*x]),x] + C*Int[(a+b*sin[e+f*x])^m*(1+Sin[e+f*x])^2,x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && EqQ[A-B+C,0] && Not[IntegerQ[2*m]]
```

```
Int[(a+b_.sin[e_.+f_.x_])^m_.*(A_.+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  (A-C)*Int[(a+b*sin[e+f*x])^m*(1+Sin[e+f*x]),x] + C*Int[(a+b*sin[e+f*x])^m*(1+Sin[e+f*x])^2,x] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A+C,0] && Not[IntegerQ[2*m]]
```

5. $\int (a+b \sin[e+f x])^m (A+B \sin[e+f x]+C \sin[e+f x]^2) dx$ when $m < -1$

1: $\int (a+b \sin[e+f x])^m (A+B \sin[e+f x]+C \sin[e+f x]^2) dx$ when $m < -1 \wedge a^2-b^2=0$

Derivation: Symmetric sine recurrence 2a with $m \rightarrow 0$ plus rule for integrands of the form $\sin[e+f x]^2 (a+b \sin[e+f x])^m$

Rule: If $m < -1 \wedge a^2-b^2=0$, then

$$\int (a+b \sin[e+f x])^m (A+B \sin[e+f x]+C \sin[e+f x]^2) dx \rightarrow$$

$$\int (a + b \sin[e + f x])^m (A + B \sin[e + f x]) dx + C \int \sin[e + f x]^2 (a + b \sin[e + f x])^m dx \rightarrow$$

$$\frac{(A b - a B + b C) \cos[e + f x] (a + b \sin[e + f x])^m}{a f (2 m + 1)} +$$

$$\frac{1}{a^2 (2 m + 1)} \int (a + b \sin[e + f x])^{m+1} (a A (m + 1) + m (b B - a C) + b C (2 m + 1) \sin[e + f x]) dx$$

Program code:

```
Int[(a_+b_.sin[e_+f_.x_])^m_*(A_+B_.sin[e_+f_.x_]+C_.sin[e_+f_.x_]^2),x_Symbol] :=
  (A*b-a*B+b*C)*Cos[e+f*x]*(a+b*sin[e+f*x])^m/(a*f*(2*m+1)) +
  1/(a^2*(2*m+1))*Int[(a+b*sin[e+f*x])^(m+1)*Simp[a*A*(m+1)+m*(b*B-a*C)+b*C*(2*m+1)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && LtQ[m,-1] && EqQ[a^2-b^2,0]
```

```
Int[(a_+b_.sin[e_+f_.x_])^m_*(A_+C_.sin[e_+f_.x_]^2),x_Symbol] :=
  b*(A+C)*Cos[e+f*x]*(a+b*sin[e+f*x])^m/(a*f*(2*m+1)) +
  1/(a^2*(2*m+1))*Int[(a+b*sin[e+f*x])^(m+1)*Simp[a*A*(m+1)-a*C*m+b*C*(2*m+1)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && LtQ[m,-1] && EqQ[a^2-b^2,0]
```

2: $\int (a + b \sin[e + f x])^m (A + B \sin[e + f x] + C \sin[e + f x]^2) dx$ when $m < -1 \wedge a^2 - b^2 \neq 0$

Derivation: Nondegenerate sine recurrence 1a with $n \rightarrow 0$, $p \rightarrow 0$

Rule: If $m < -1 \wedge a^2 - b^2 \neq 0$, then

$$\int (a + b \sin[e + f x])^m (A + B \sin[e + f x] + C \sin[e + f x]^2) dx \rightarrow$$

$$- \frac{(A b^2 - a b B + a^2 C) \cos[e + f x] (a + b \sin[e + f x])^{m+1}}{b f (m+1) (a^2 - b^2)} +$$

$$\frac{1}{b (m+1) (a^2 - b^2)} \int (a + b \sin[e + f x])^{m+1} (b (a A - b B + a C) (m+1) - (A b^2 - a b B + a^2 C + b (A b - a B + b C) (m+1)) \sin[e + f x]) dx$$

Program code:

```
Int[(a_.+b_.sin[e_.+f_.x_])^m_*(A_.+B_.sin[e_.+f_.x_]+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  -(A*b^2-a*b*B+a^2*C)*Cos[e+f*x]*(a+b*sin[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) +
  1/(b*(m+1)*(a^2-b^2))*
  Int[(a+b*sin[e+f*x])^(m+1)*Simp[b*(a*A-b*B+a*C)*(m+1)-(A*b^2-a*b*B+a^2*C+b*(A*b-a*B+b*C)*(m+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && LtQ[m,-1] && NeQ[a^2-b^2,0]
```

```
Int[(a+b_.sin[e_.+f_.x_])^m_*(A_.+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  -(A*b^2+a^2*C)*Cos[e+f*x]*(a+b*sin[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) +
  1/(b*(m+1)*(a^2-b^2))*
  Int[(a+b*sin[e+f*x])^(m+1)*Simp[a*b*(A+C)*(m+1)-(A*b^2+a^2*C+b^2*(A+C)*(m+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && LtQ[m,-1] && NeQ[a^2-b^2,0]
```


6: $\int (a + b \sin[e + f x])^m (A + B \sin[e + f x] + C \sin[e + f x]^2) dx$ when $m \notin \mathbb{Z}$

Derivation: Nondegenerate sine recurrence 1b with $m \rightarrow 0$, $p \rightarrow 0$

Rule: If $m \notin \mathbb{Z}$, then

$$\int (a + b \sin[e + f x])^m (A + B \sin[e + f x] + C \sin[e + f x]^2) dx \rightarrow$$

$$-\frac{C \cos[e + f x] (a + b \sin[e + f x])^{m+1}}{b f (m+2)} + \frac{1}{b f (m+2)} \int (a + b \sin[e + f x])^m (A b (m+2) + b C (m+1) + (b B (m+2) - a C) \sin[e + f x]) dx$$

Program code:

```
Int[(a_+b_.sin[e_+f_.x_])^m_.*(A_+B_.sin[e_+f_.x_]+C_.sin[e_+f_.x_]^2),x_Symbol] :=
  -C*cos[e+f*x]*(a+b*sin[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[(a+b*sin[e+f*x])^m*Simp[A*b*(m+2)+b*C*(m+1)+(b*B*(m+2)-a*C)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && Not[LtQ[m,-1]]
```

```
Int[(a_+b_.sin[e_+f_.x_])^m_.*(A_+C_.sin[e_+f_.x_]^2),x_Symbol] :=
  -C*cos[e+f*x]*(a+b*sin[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[(a+b*sin[e+f*x])^m*Simp[A*b*(m+2)+b*C*(m+1)-a*C*Sine[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && Not[LtQ[m,-1]]
```

Rules for integrands of the form $(b \sin[e + f x]^p)^m (A + B \sin[e + f x] + C \sin[e + f x]^2)$

1: $\int (b \sin[e + f x]^p)^m (A + B \sin[e + f x] + C \sin[e + f x]^2) dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b \sin[e+fx]^p)^m}{(b \sin[e+fx])^{mp}} == 0$

Rule: If $m \notin \mathbb{Z}$, then

$$\int (b \sin[e+fx]^p)^m (A+B \sin[e+fx]+C \sin[e+fx]^2) dx \rightarrow \frac{(b \sin[e+fx]^p)^m}{(b \sin[e+fx])^{mp}} \int (b \sin[e+fx])^{mp} (A+B \sin[e+fx]+C \sin[e+fx]^2) dx$$

Program code:

```
Int[(b_.sin[e_.+f_.x_]^p_)^m_*(A_.+B_.sin[e_.+f_.x_]+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  (b*sin[e+f*x]^p)^m/(b*sin[e+f*x])^(m*p)*Int[(b*sin[e+f*x])^(m*p)*(A+B*sin[e+f*x]+C*sin[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,B,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.cos[e_.+f_.x_]^p_)^m_*(A_.+B_.cos[e_.+f_.x_]+C_.cos[e_.+f_.x_]^2),x_Symbol] :=
  (b*cos[e+f*x]^p)^m/(b*cos[e+f*x])^(m*p)*Int[(b*cos[e+f*x])^(m*p)*(A+B*cos[e+f*x]+C*cos[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,B,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.sin[e_.+f_.x_]^p_)^m_*(A_.+C_.sin[e_.+f_.x_]^2),x_Symbol] :=
  (b*sin[e+f*x]^p)^m/(b*sin[e+f*x])^(m*p)*Int[(b*sin[e+f*x])^(m*p)*(A+C*sin[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.cos[e_.+f_.x_]^p_)^m_*(A_.+C_.cos[e_.+f_.x_]^2),x_Symbol] :=
  (b*cos[e+f*x]^p)^m/(b*cos[e+f*x])^(m*p)*Int[(b*cos[e+f*x])^(m*p)*(A+C*cos[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,C,m,p},x] && Not[IntegerQ[m]]
```