Rules for integrands of the form $(a\,\text{Trg}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n$

1. $\int (a\,\text{Sin}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx$

   1: $\int (a\,\text{Sin}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx$ when $m+n-1==0$

Rule: If $m+n-1==0$, then

$$\int (a\,\text{Sin}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx \longrightarrow -\frac{b\,(a\,\text{Sin}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^{n-1}}{f\,m}$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  -b*(a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*m) /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m+n-1,0]
```

2: $\int \text{Sin}[e+f\,x]^m\,\text{Tan}[e+f\,x]^n\,dx$ when $\left(m \mid n \mid \frac{m+n-1}{2}\right) \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\left(m \mid n \mid \frac{m+n-1}{2}\right) \in \mathbb{Z}$, then

$$\text{Sin}[e+f\,x]^m\,\text{Tan}[e+f\,x]^n = -\frac{1}{f}\,\text{Subst}\left[\frac{\left(1-x^2\right)^{\frac{m+n-1}{2}}}{x^n},\,x,\,\text{Cos}[e+f\,x]\right]\partial_x\text{Cos}[e+f\,x]$$

Rule: If $\left(m \mid n \mid \frac{m+n-1}{2}\right) \in \mathbb{Z}$, then

$$\int \text{Sin}[e+f\,x]^m\,\text{Tan}[e+f\,x]^n\,dx \longrightarrow -\frac{1}{f}\,\text{Subst}\left[\int \frac{\left(1-x^2\right)^{\frac{m+n-1}{2}}}{x^n}\,dx,\,x,\,\text{Cos}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*tan[e_.+f_.*x_]^n_.,x_Symbol] :=
  -1/f*Subst[Int[(1-x^2)^((m+n-1)/2)/x^n,x],x,Cos[e+f*x]] /;
FreeQ[{e,f},x] && IntegersQ[m,n,(m+n-1)/2]
```

3: $\int \text{Sin}[e+f\,x]^m\,(b\,\text{Tan}[e+f\,x])^n\,dx$ when $\frac{m}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2 = \frac{\text{Tan}[z]^2}{1+\text{Tan}[z]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\text{Sin}[e+f\,x]^m\,F[b\,\text{Tan}[e+f\,x]] = \frac{b}{f}\,\text{Subst}\left[\frac{x^m\,F[x]}{\left(b^2+x^2\right)^{\frac{m}{2}+1}},\,x,\,b\,\text{Tan}[e+f\,x]\right]\partial_x\,(b\,\text{Tan}[e+f\,x])$$

Rule: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\int \text{Sin}[e+f\,x]^m\,(b\,\text{Tan}[e+f\,x])^n\,dx \;\to\; \frac{b}{f}\,\text{Subst}\Big[\int \frac{x^{m+n}}{(b^2+x^2)^{\frac{m}{2}+1}}\,dx,\;x,\;b\,\text{Tan}[e+f\,x]\Big]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  b*ff/f*Subst[Int[(ff*x)^(m+n)/(b^2+ff^2*x^2)^(m/2+1),x],x,b*Tan[e+f*x]/ff]] /;
FreeQ[{b,e,f,n},x] && IntegerQ[m/2]
```

4: $\int (a\,\text{Sin}[e+f\,x])^m\,\text{Tan}[e+f\,x]^n\,dx$ when $\frac{n+1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n+1}{2}\in\mathbb{Z}$, then $\text{Tan}[e+f\,x]^n\,F[a\,\text{Sin}[e+f\,x]] == \frac{1}{f}\,\text{Subst}\Big[\frac{x^n\,F[x]}{(a^2-x^2)^{\frac{n+1}{2}}},\;x,\;a\,\text{Sin}[e+f\,x]\Big]\,\partial_x\,(a\,\text{Sin}[e+f\,x])$

Rule: If $\frac{n+1}{2}\in\mathbb{Z}$, then

$$\int (a\,\text{Sin}[e+f\,x])^m\,\text{Tan}[e+f\,x]^n\,dx \;\to\; \frac{1}{f}\,\text{Subst}\Big[\int \frac{x^{m+n}}{(a^2-x^2)^{\frac{n+1}{2}}}\,dx,\;x,\;a\,\text{Sin}[e+f\,x]\Big]$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*tan[e_.+f_.*x_]^n_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[(ff*x)^(m+n)/(a^2-ff^2*x^2)^((n+1)/2),x],x,a*Sin[e+f*x]/ff]] /;
FreeQ[{a,e,f,m},x] && IntegerQ[(n+1)/2]
```

5. $\int (a\,\text{Sin}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx$ when $n>1$

1: $\int (a\,\text{Sin}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx$ when $n>1\land m<-1$

Reference: G&R 2.510.6, CRC 334b

Reference: G&R 2.510.3, CRC 334a

Rule: If $n>1\land m<-1$, then

$$\int (a\,\text{Sin}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx\to\frac{b\,(a\,\text{Sin}[e+f\,x])^{m+2}\,(b\,\text{Tan}[e+f\,x])^{n-1}}{a^2\,f\,(n-1)}-\frac{b^2\,(m+2)}{a^2\,(n-1)}\int (a\,\text{Sin}[e+f\,x])^{m+2}\,(b\,\text{Tan}[e+f\,x])^{n-2}\,dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Sin[e+f*x])^(m+2)*(b*Tan[e+f*x])^(n-1)/(a^2*f*(n-1)) -
  b^2*(m+2)/(a^2*(n-1))*Int[(a*Sin[e+f*x])^(m+2)*(b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[n,1] && (LtQ[m,-1] || EqQ[m,-1] && EqQ[n,3/2]) && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n > 1$

Reference: G&R 2.510.1

Reference: G&R 2.510.4

Rule: If $n > 1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow \frac{b (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n-1}}{f (n-1)} - \frac{b^2 (m+n-1)}{n-1} \int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n-2} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*(n-1)) -
  b^2*(m+n-1)/(n-1)*Int[(a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f,m},x] && GtQ[n,1] && IntegersQ[2*m,2*n] && Not[GtQ[m,1] && Not[IntegerQ[(m-1)/2]]]
```

6. $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n < -1$

1: $\int \frac{\sqrt{a \operatorname{Sin}[e+f x]}}{(b \operatorname{Tan}[e+f x])^{3/2}} \, dx$

Rule:

$$\int \frac{\sqrt{a \operatorname{Sin}[e+f x]}}{(b \operatorname{Tan}[e+f x])^{3/2}} \, dx \to \frac{2 \sqrt{a \operatorname{Sin}[e+f x]}}{b f \sqrt{b \operatorname{Tan}[e+f x]}} + \frac{a^2}{b^2} \int \frac{\sqrt{b \operatorname{Tan}[e+f x]}}{(a \operatorname{Sin}[e+f x])^{3/2}} \, dx$$

Program code:

```
Int[Sqrt[a_.*sin[e_.+f_.*x_]]/(b_.*tan[e_.+f_.*x_])^(3/2),x_Symbol]:=
  2*Sqrt[a*Sin[e+f*x]]/(b*f*Sqrt[b*Tan[e+f*x]]) + a^2/b^2*Int[Sqrt[b*Tan[e+f*x]]/(a*Sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,e,f},x]
```

2: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n < -1 \land m > 1$

Reference: G&R 2.510.5, CRC 323a

Reference: G&R 2.510.2, CRC 323b

Rule: If $n < -1 \land m > 1$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow \frac{(a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1}}{b f m} - \frac{a^2 (n+1)}{b^2 m} \int (a \operatorname{Sin}[e+f x])^{m-2} (b \operatorname{Tan}[e+f x])^{n+2} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*m) -
  a^2*(n+1)/(b^2*m)*Int[(a*Sin[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && LtQ[n,-1] && GtQ[m,1] && IntegersQ[2*m,2*n]
```

3: $\int (a \sin[e + f x])^m (b \tan[e + f x])^n \, dx$ when $n < -1 \land m + n + 1 \neq 0$

Reference: G&R 2.510.4

Reference: G&R 2.510.1

Rule: If $n < -1 \land m + n + 1 \neq 0$, then

$$\int (a \sin[e + f x])^m (b \tan[e + f x])^n \, dx \longrightarrow \frac{(a \sin[e + f x])^m (b \tan[e + f x])^{n+1}}{b f (m + n + 1)} - \frac{n + 1}{b^2 (m + n + 1)} \int (a \sin[e + f x])^m (b \tan[e + f x])^{n+2} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  (a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*(m+n+1)) -
  (n+1)/(b^2*(m+n+1))*Int[(a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f,m},x] && LtQ[n,-1] && NeQ[m+n+1,0] && IntegersQ[2*m,2*n] && Not[EqQ[n,-3/2] && EqQ[m,1]]
```

7: $\int (a \operatorname{Sin}[e + f x])^m (b \operatorname{Tan}[e + f x])^n \, dx$ when $m > 1$

Reference: G&R 2.510.2, CRC 323b

Reference: G&R 2.510.5, CRC 323a

Rule: If $m > 1$, then

$$\int (a \operatorname{Sin}[e + f x])^m (b \operatorname{Tan}[e + f x])^n \, dx \longrightarrow$$
$$-\frac{b (a \operatorname{Sin}[e + f x])^m (b \operatorname{Tan}[e + f x])^{n-1}}{f m} + \frac{a^2 (m + n - 1)}{m} \int (a \operatorname{Sin}[e + f x])^{m-2} (b \operatorname{Tan}[e + f x])^n \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_.,x_Symbol]:=
  -b*(a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*m) +
  a^2*(m+n-1)/m*Int[(a*Sin[e+f*x])^(m-2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && (GtQ[m,1] || EqQ[m,1] && EqQ[n,1/2]) && IntegersQ[2*m,2*n]
```

8: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $m<-1 \land m+n+1 \neq 0$

Reference: G&R 2.510.3, CRC 334a

Reference: G&R 2.510.6, CRC 334b

Rule: If $m<-1 \land m+n+1 \neq 0$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \to \frac{b (a \operatorname{Sin}[e+f x])^{m+2} (b \operatorname{Tan}[e+f x])^{n-1}}{a^2 f (m+n+1)} + \frac{m+2}{a^2 (m+n+1)} \int (a \operatorname{Sin}[e+f x])^{m+2} (b \operatorname{Tan}[e+f x])^n \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_.,x_Symbol]:=
  b*(a*Sin[e+f*x])^(m+2)*(b*Tan[e+f*x])^(n-1)/(a^2*f*(m+n+1)) +
  (m+2)/(a^2*(m+n+1))*Int[(a*Sin[e+f*x])^(m+2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && LtQ[m,-1] && NeQ[m+n+1,0] && IntegersQ[2*m,2*n]
```

9: $\int (a\,\mathrm{Sin}[e+f\,x])^m\,\mathrm{Tan}[e+f\,x]^n\,dx$ when $n \in \mathbb{Z} \land m \notin \mathbb{Z}$

Derivation: Algebraic normalization

Basis: $\mathrm{Tan}[z] == \frac{\mathrm{Sin}[z]}{\mathrm{Cos}[z]}$

Rule: If $n \in \mathbb{Z} \land m \notin \mathbb{Z}$, then

$$\int (a\,\mathrm{Sin}[e+f\,x])^m\,\mathrm{Tan}[e+f\,x]^n\,dx \longrightarrow \frac{1}{a^n}\int \frac{(a\,\mathrm{Sin}[e+f\,x])^{m+n}}{\mathrm{Cos}[e+f\,x]^n}\,dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*tan[e_.+f_.*x_]^n_,x_Symbol]:=
  1/a^n*Int[(a*Sin[e+f*x])^(m+n)/Cos[e+f*x]^n,x] /;
FreeQ[{a,e,f,m},x] && IntegerQ[n] && Not[IntegerQ[m]]
```

10. $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n \notin \mathbb{Z}$

1: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n \notin \mathbb{Z} \wedge m < 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(\operatorname{Cos}[e+f x])^n (b \operatorname{Tan}[e+f x])^n}{(a \operatorname{Sin}[e+f x])^n} == 0$

Rule: If $n \notin \mathbb{Z} \wedge m < 0$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow \frac{(\operatorname{Cos}[e+f x])^n (b \operatorname{Tan}[e+f x])^n}{(a \operatorname{Sin}[e+f x])^n} \int \frac{(a \operatorname{Sin}[e+f x])^{m+n}}{\operatorname{Cos}[e+f x]^n} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  Cos[e+f*x]^n*(b*Tan[e+f*x])^n/(a*Sin[e+f*x])^n*Int[(a*Sin[e+f*x])^(m+n)/Cos[e+f*x]^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[n]] && (ILtQ[m,0] || EqQ[m,1] && EqQ[n,-1/2] || IntegersQ[m-1/2,n-1/2])
```

2: $\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{(\operatorname{Cos}[e+f x])^n (b \operatorname{Tan}[e+f x])^n}{(a \operatorname{Sin}[e+f x])^n} == 0$

Rule: If $n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Sin}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \rightarrow \frac{a (\operatorname{Cos}[e+f x])^{n+1} (b \operatorname{Tan}[e+f x])^{n+1}}{b (a \operatorname{Sin}[e+f x])^{n+1}} \int \frac{(a \operatorname{Sin}[e+f x])^{m+n}}{\operatorname{Cos}[e+f x]^n} \, dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  a*Cos[e+f*x]^(n+1)*(b*Tan[e+f*x])^(n+1)/(b*(a*Sin[e+f*x])^(n+1))*Int[(a*Sin[e+f*x])^(m+n)/Cos[e+f*x]^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[n]]
```

2: $\int (a \operatorname{Cos}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

–

Basis: $\partial_x \left((a \operatorname{Cos}[e+f x])^m \left(\frac{\operatorname{Sec}[e+f x]}{a} \right)^m \right) == 0$

Rule: If $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Cos}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \to (a \operatorname{Cos}[e+f x])^{\operatorname{FracPart}[m]} \left(\frac{\operatorname{Sec}[e+f x]}{a} \right)^{\operatorname{FracPart}[m]} \int \frac{(b \operatorname{Tan}[e+f x])^n}{\left(\frac{\operatorname{Sec}[e+f x]}{a} \right)^m} \, dx$$

Program code:

```
Int[(a_.*cos[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Cos[e+f*x])^FracPart[m]*(Sec[e+f*x]/a)^FracPart[m]*Int[(b*Tan[e+f*x])^n/(Sec[e+f*x]/a)^m,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

3: $\int (a \operatorname{Cot}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((a \operatorname{Cot}[e+f x])^m (b \operatorname{Tan}[e+f x])^m\right) == 0$

Rule: If $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Cot}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow (a \operatorname{Cot}[e+f x])^m (b \operatorname{Tan}[e+f x])^m \int (b \operatorname{Tan}[e+f x])^{n-m} \, dx$$

Program code:

```
Int[(a_.*cot[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Cot[e+f*x])^m*(b*Tan[e+f*x])^m*Int[(b*Tan[e+f*x])^(n-m),x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

4. $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$

1: $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $m+n+1 == 0$

Rule: If $m+n+1 == 0$, then

$$\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow -\frac{(a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1}}{b f m}$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  -(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*m) /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m+n+1,0]
```

2: $\int (a\,\mathrm{Sec}[e+f\,x])^m\,\mathrm{Tan}[e+f\,x]^n\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}\land\neg\left(\frac{m}{2}\in\mathbb{Z}\land 0<m<n+1\right)$

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then

$$\mathrm{Tan}[e+f\,x]^n\,F[\mathrm{Sec}[e+f\,x]]=\frac{1}{f}\,\mathrm{Subst}\left[\frac{F[x]\,\left(-1+x^2\right)^{\frac{n-1}{2}}}{x},\,x,\,\mathrm{Sec}[e+f\,x]\right]\partial_x\mathrm{Sec}[e+f\,x]$$

Rule: If $\frac{n-1}{2}\in\mathbb{Z}\land\neg\left(\frac{m}{2}\in\mathbb{Z}\land 0<m<n+1\right)$, then

$$\int (a\,\mathrm{Sec}[e+f\,x])^m\,\mathrm{Tan}[e+f\,x]^n\,dx\,\longrightarrow\,\frac{a}{f}\,\mathrm{Subst}\left[\int (a\,x)^{m-1}\,\left(-1+x^2\right)^{\frac{n-1}{2}}\,dx,\,x,\,\mathrm{Sec}[e+f\,x]\right]$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  a/f*Subst[Int[(a*x)^(m-1)*(-1+x^2)^((n-1)/2),x],x,Sec[e+f*x]] /;
FreeQ[{a,e,f,m},x] && IntegerQ[(n-1)/2] && Not[IntegerQ[m/2] && LtQ[0,m,n+1]]
```

3: $\int \mathrm{Sec}[e+f\,x]^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $\frac{m}{2}\in\mathbb{Z}\land\neg\left(\frac{n-1}{2}\in\mathbb{Z}\land 0<n<m-1\right)$

Derivation: Integration by substitution

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\mathrm{Sec}[e+f\,x]^m\,F[\mathrm{Tan}[e+f\,x]]=\frac{1}{f}\,\mathrm{Subst}\left[F[x]\,\left(1+x^2\right)^{\frac{m}{2}-1},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\mathrm{Tan}[e+f\,x]$$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land\neg\left(\frac{n-1}{2}\in\mathbb{Z}\land 0<n<m-1\right)$, then

$$\int \mathrm{Sec}[e+f\,x]^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx\,\rightarrow\,\frac{1}{f}\,\mathrm{Subst}\left[\int (b\,x)^n\,\left(1+x^2\right)^{\frac{m}{2}-1}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_*(b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  1/f*Subst[Int[(b*x)^n*(1+x^2)^(m/2-1),x],x,Tan[e+f*x]] /;
FreeQ[{b,e,f,n},x] && IntegerQ[m/2] && Not[IntegerQ[(n-1)/2] && LtQ[0,n,m-1]]
```

4. $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n < -1$

1: $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n < -1 \land \left(m > 1 \lor m == 1 \land n == -\frac{3}{2}\right)$

Reference: G&R 2.510.5, CRC 323a

Reference: G&R 2.510.2, CRC 323b

Rule: If $n < -1 \land \left(m > 1 \lor m == 1 \land n == -\frac{3}{2}\right)$, then

$$\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \to \frac{a^2 (a \operatorname{Sec}[e+f x])^{m-2} (b \operatorname{Tan}[e+f x])^{n+1}}{b f (n+1)} - \frac{a^2 (m-2)}{b^2 (n+1)} \int (a \operatorname{Sec}[e+f x])^{m-2} (b \operatorname{Tan}[e+f x])^{n+2} \, dx$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+1)/(b*f*(n+1)) -
  a^2*(m-2)/(b^2*(n+1))*Int[(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && LtQ[n,-1] && (GtQ[m,1] || EqQ[m,1] && EqQ[n,-3/2]) && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n < -1$

Reference: G&R 2.510.4

Reference: G&R 2.510.1

Rule: If $n < -1$, then

$$\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow$$
$$\frac{(a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1}}{b f (n+1)} - \frac{m+n+1}{b^2 (n+1)} \int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+2} \, dx$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*(n+1)) -
  (m+n+1)/(b^2*(n+1))*Int[(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f,m},x] && LtQ[n,-1] && IntegersQ[2*m,2*n]
```

5. $\int (a\,\text{Sec}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx$ when $n>1$

1: $\int (a\,\text{Sec}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx$ when $n>1\land\left(m<-1\lor m==-1\land n==\frac{3}{2}\right)$

Reference: G&R 2.510.6, CRC 334b

Reference: G&R 2.510.3, CRC 334a

Rule: If $n>1\land\left(m<-1\lor m==-1\land n==\frac{3}{2}\right)$, then

$$\int (a\,\text{Sec}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^n\,dx\longrightarrow\frac{b\,(a\,\text{Sec}[e+f\,x])^m\,(b\,\text{Tan}[e+f\,x])^{n-1}}{f\,m}-\frac{b^2\,(n-1)}{a^2\,m}\int (a\,\text{Sec}[e+f\,x])^{m+2}\,(b\,\text{Tan}[e+f\,x])^{n-2}\,dx$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*m) -
  b^2*(n-1)/(a^2*m)*Int[(a*Sec[e+f*x])^(m+2)*(b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[n,1] && (LtQ[m,-1] || EqQ[m,-1] && EqQ[n,3/2]) && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n > 1 \land m+n-1 \neq 0$

Reference: G&R 2.510.1

Reference: G&R 2.510.4

Rule: If $n > 1 \land m+n-1 \neq 0$, then

$$\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow$$
$$\frac{b (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n-1}}{f (m+n-1)} - \frac{b^2 (n-1)}{m+n-1} \int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n-2} \, dx$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*(m+n-1)) -
  b^2*(n-1)/(m+n-1)*Int[(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f,m},x] && GtQ[n,1] && NeQ[m+n-1,0] && IntegersQ[2*m,2*n]
```

6: $\int (a \operatorname{Sec}[e + f x])^m (b \operatorname{Tan}[e + f x])^n \, dx$ when $m < -1$

Reference: G&R 2.510.3, CRC 334a

Reference: G&R 2.510.6, CRC 334b

Rule: If $m < -1$, then

$$\int (a \operatorname{Sec}[e + f x])^m (b \operatorname{Tan}[e + f x])^n \, dx \longrightarrow -\frac{(a \operatorname{Sec}[e + f x])^m (b \operatorname{Tan}[e + f x])^{n+1}}{b f m} + \frac{m + n + 1}{a^2 m} \int (a \operatorname{Sec}[e + f x])^{m+2} (b \operatorname{Tan}[e + f x])^n \, dx$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  -(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*m) +
  (m+n+1)/(a^2*m)*Int[(a*Sec[e+f*x])^(m+2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && (LtQ[m,-1] || EqQ[m,-1] && EqQ[n,-1/2]) && IntegersQ[2*m,2*n]
```

7: $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $m>1 \land m+n-1 \neq 0$

Reference: G&R 2.510.2, CRC 323b

Reference: G&R 2.510.5, CRC 323a

Rule: If $m>1 \land m+n-1 \neq 0$, then

$$\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \longrightarrow$$

$$\frac{a^2 (a \operatorname{Sec}[e+f x])^{m-2} (b \operatorname{Tan}[e+f x])^{n+1}}{b f (m+n-1)}+\frac{a^2 (m-2)}{(m+n-1)} \int (a \operatorname{Sec}[e+f x])^{m-2} (b \operatorname{Tan}[e+f x])^n \, dx$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+1)/(b*f*(m+n-1)) +
  a^2*(m-2)/(m+n-1)*Int[(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && (GtQ[m,1] || EqQ[m,1] && EqQ[n,1/2]) && NeQ[m+n-1,0] && IntegersQ[2*m,2*n]
```

8: $\int \frac{\operatorname{Sec}[e+f x]}{\sqrt{b \operatorname{Tan}[e+f x]}} \, dx$

Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{\sqrt{\operatorname{Sin}[e+f x]}}{\sqrt{\operatorname{Cos}[e+f x]} \sqrt{b \operatorname{Tan}[e+f x]}} == 0$

Rule:

$$\int \frac{\text{Sec}[e+f x]}{\sqrt{b\,\text{Tan}[e+f x]}}\,dx \to \frac{\sqrt{\text{Sin}[e+f x]}}{\sqrt{\text{Cos}[e+f x]}\,\sqrt{b\,\text{Tan}[e+f x]}}\int \frac{1}{\sqrt{\text{Cos}[e+f x]}\,\sqrt{\text{Sin}[e+f x]}}\,dx$$

Program code:

```
Int[sec[e_.+f_.*x_]/Sqrt[b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  Sqrt[Sin[e+f*x]]/(Sqrt[Cos[e+f*x]]*Sqrt[b*Tan[e+f*x]])*Int[1/(Sqrt[Cos[e+f*x]]*Sqrt[Sin[e+f*x]]),x] /;
FreeQ[{b,e,f},x]
```

9: $\int \frac{\sqrt{b\,\text{Tan}[e+f x]}}{\text{Sec}[e+f x]}\,dx$

Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{\sqrt{\text{Cos}[e+f x]}\,\sqrt{b\,\text{Tan}[e+f x]}}{\sqrt{\text{Sin}[e+f x]}} == 0$

Rule:

$$\int \frac{\sqrt{b\,\text{Tan}[e+f x]}}{\text{Sec}[e+f x]}\,dx \to \frac{\sqrt{\text{Cos}[e+f x]}\,\sqrt{b\,\text{Tan}[e+f x]}}{\sqrt{\text{Sin}[e+f x]}}\int \sqrt{\text{Cos}[e+f x]}\,\sqrt{\text{Sin}[e+f x]}\,dx$$

Program code:

```
Int[Sqrt[b_.*tan[e_.+f_.*x_]]/sec[e_.+f_.*x_],x_Symbol] :=
  Sqrt[Cos[e+f*x]]*Sqrt[b*Tan[e+f*x]]/Sqrt[Sin[e+f*x]]*Int[Sqrt[Cos[e+f*x]]*Sqrt[Sin[e+f*x]],x] /;
FreeQ[{b,e,f},x]
```

10: $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $n+\frac{1}{2} \in \mathbb{Z} \wedge m+\frac{1}{2} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{(b \operatorname{Tan}[e+f x])^n}{(a \operatorname{Sec}[e+f x])^n (b \operatorname{Sin}[e+f x])^n} == 0$

Rule: If $n+\frac{1}{2} \in \mathbb{Z} \wedge m+\frac{1}{2} \in \mathbb{Z}$, then

$$\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \to \frac{a^{m+n} (b \operatorname{Tan}[e+f x])^n}{(a \operatorname{Sec}[e+f x])^n (b \operatorname{Sin}[e+f x])^n} \int \frac{(b \operatorname{Sin}[e+f x])^n}{\operatorname{Cos}[e+f x]^{m+n}} \, dx$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a^(m+n)*(b*Tan[e+f*x])^n/((a*Sec[e+f*x])^n*(b*Sin[e+f*x])^n)*Int[(b*Sin[e+f*x])^n/Cos[e+f*x]^(m+n),x] /;
FreeQ[{a,b,e,f,m,n},x] && IntegerQ[n+1/2] && IntegerQ[m+1/2]
```

11: $\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx$ when $\frac{n-1}{2} \notin \mathbb{Z} \wedge \frac{m}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{(a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1} \left(\operatorname{Cos}[e+f x]^2\right)^{\frac{m+n+1}{2}}}{(b \operatorname{Sin}[e+f x])^{n+1}} == 0$

Basis: $\operatorname{Cos}[e+f x] \, F[\operatorname{Sin}[e+f x]] == \frac{1}{b f} \operatorname{Subst}\left[F\left[\frac{x}{b}\right], x, b \operatorname{Sin}[e+f x]\right] \partial_x (b \operatorname{Sin}[e+f x])$

Note: If $\frac{n}{2} \in \mathbb{Z}$, then $\frac{(a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1} \left(\operatorname{Cos}[e+f x]^2\right)^{\frac{m+n+1}{2}}}{(b \operatorname{Sin}[e+f x])^{n+1}} == (a \operatorname{Sec}[e+f x])^{m+1} \left(\operatorname{Cos}[e+f x]^2\right)^{\frac{m+1}{2}}$

Note: If $\frac{n}{2} \in \mathbb{Z}$ and m is a third-integer integration of $\frac{x^n}{\left(1-\frac{x^2}{b^2}\right)^{\frac{m+n+1}{2}}}$ results in a complicated antiderivative involving elliptic integrals and the imaginary unit.

Rule: If $\frac{n-1}{2} \notin \mathbb{Z} \land \frac{m}{2} \notin \mathbb{Z}$, then

$$\int (a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^n \, dx \to \frac{(a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1} \left(\operatorname{Cos}[e+f x]^2\right)^{\frac{m+n+1}{2}}}{(b \operatorname{Sin}[e+f x])^{n+1}} \int \frac{\operatorname{Cos}[e+f x] (b \operatorname{Sin}[e+f x])^n}{\left(1-\operatorname{Sin}[e+f x]^2\right)^{\frac{m+n+1}{2}}} \, dx$$

$$\to \frac{(a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1} \left(\operatorname{Cos}[e+f x]^2\right)^{\frac{m+n+1}{2}}}{b f (b \operatorname{Sin}[e+f x])^{n+1}} \operatorname{Subst}\left[\int \frac{x^n}{\left(1-\frac{x^2}{b^2}\right)^{\frac{m+n+1}{2}}} \, dx, x, b \operatorname{Sin}[e+f x]\right]$$

$$\to \frac{(a \operatorname{Sec}[e+f x])^m (b \operatorname{Tan}[e+f x])^{n+1} \left(\operatorname{Cos}[e+f x]^2\right)^{\frac{m+n+1}{2}}}{b f (n+1)} \operatorname{Hypergeometric2F1}\left[\frac{n+1}{2}, \frac{m+n+1}{2}, \frac{n+3}{2}, \operatorname{Sin}[e+f x]^2\right]$$

Program code:

```
(* Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  (a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)*(Cos[e+f*x]^2)^((m+n+1)/2)/(b*f*(b*Sin[e+f*x])^(n+1))*
    Subst[Int[x^n/(1-x^2/b^2)^((m+n+1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[(n-1)/2]] && Not[IntegerQ[m/2]] *)
```

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  (a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)*(Cos[e+f*x]^2)^((m+n+1)/2)/(b*f*(n+1))*
    Hypergeometric2F1[(n+1)/2,(m+n+1)/2,(n+3)/2,Sin[e+f*x]^2] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[(n-1)/2]] && Not[IntegerQ[m/2]]
```

5: $\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((a\,\mathrm{Csc}[e+f\,x])^m\,(a\,\mathrm{Sin}[e+f\,x])^m\right) == 0$

Rule: If $m \notin \mathbb{Z} \land n \notin \mathbb{Z}$, then

$$\int (a\,\mathrm{Csc}[e+f\,x])^m\,(b\,\mathrm{Tan}[e+f\,x])^n\,dx \longrightarrow (a\,\mathrm{Csc}[e+f\,x])^{\mathrm{FracPart}[m]}\left(\frac{\mathrm{Sin}[e+f\,x]}{a}\right)^{\mathrm{FracPart}[m]}\int \frac{(b\,\mathrm{Tan}[e+f\,x])^n}{\left(\frac{\mathrm{Sin}[e+f\,x]}{a}\right)^m}\,dx$$

Program code:

```
Int[(a_.*csc[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Csc[e+f*x])^FracPart[m]*(Sin[e+f*x]/a)^FracPart[m]*Int[(b*Tan[e+f*x])^n/(Sin[e+f*x]/a)^m,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```