Rules for integrands of the form $(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+A\,\mathrm{Tan}[e+f\,x]^2\right)\,dx$

Derivation: Integration by substitution

Basis: $F[\mathrm{Tan}[e+f\,x]]\,\left(A+A\,\mathrm{Tan}[e+f\,x]^2\right) = \frac{A}{f}\,\mathrm{Subst}[F[x],\,x,\,\mathrm{Tan}[e+f\,x]]\,\partial_x\mathrm{Tan}[e+f\,x]$

Rule:

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+A\,\mathrm{Tan}[e+f\,x]^2\right)\,dx \longrightarrow \frac{A}{f}\,\mathrm{Subst}\left[\int (a+b\,x)^m\,(c+d\,x)^n\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])^n_.*(A_+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  A/f*Subst[Int[(a+b*x)^m*(c+d*x)^n,x],x,Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && EqQ[A,C]
```

2. $\int (a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,c^2+d^2\neq 0$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,c^2+d^2\neq 0\,\land\,n<-1$

Derivation: Algebraic expansion, nondegenerate tangent recurrence 1c with $c\to 1,\ d\to 0,\ A\to c,\ B\to d,\ C\to 0,\ n\to 0,\ p\to 0$ and algebraic simplification

Basis: $A+B\,z+C\,z^2 = \frac{c^2\,C-B\,c\,d+A\,d^2}{d^2} - \frac{(c+d\,z)\,(c\,C-B\,d-C\,d\,z)}{d^2}$

Rule: If $b\,c-a\,d\neq 0\,\land\,c^2+d^2\neq 0\,\land\,n<-1$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)\,dx \longrightarrow$$

$$\frac{c^2\,C-B\,c\,d+A\,d^2}{d^2}\int (a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx - \frac{1}{d^2}\int (a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}\,(c\,C-B\,d-C\,d\,\mathrm{Tan}[e+f\,x])\,dx \longrightarrow$$

$$-\frac{(b\,c-a\,d)\left(c^2\,C-B\,c\,d+A\,d^2\right)(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d^2\,f\,(n+1)\left(c^2+d^2\right)}+\frac{1}{d\left(c^2+d^2\right)}\int(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}\cdot$$
$$\left(a\,d\,(A\,c-c\,C+B\,d)+b\left(c^2\,C-B\,c\,d+A\,d^2\right)+d\,(A\,b\,c+a\,B\,c-b\,c\,C-a\,A\,d+b\,B\,d+a\,C\,d)\,\mathrm{Tan}[e+f\,x]+b\,C\left(c^2+d^2\right)\mathrm{Tan}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  -(b*c-a*d)*(c^2*C-B*c*d+A*d^2)*(c+d*Tan[e+f*x])^(n+1)/(d^2*f*(n+1)*(c^2+d^2)) +
  1/(d*(c^2+d^2))*Int[(c+d*Tan[e+f*x])^(n+1)*
    Simp[a*d*(A*c-c*C+B*d)+b*(c^2*C-B*c*d+A*d^2)+d*(A*b*c+a*B*c-b*c*C-a*A*d+b*B*d+a*C*d)*Tan[e+f*x]+b*C*(c^2+d^2)*Tan[e+f*x]^2,x],x]
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[c^2+d^2,0] && LtQ[n,-1]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  -(b*c-a*d)*(c^2*C+A*d^2)*(c+d*Tan[e+f*x])^(n+1)/(d^2*f*(n+1)*(c^2+d^2)) +
  1/(d*(c^2+d^2))*Int[(c+d*Tan[e+f*x])^(n+1)*
    Simp[a*d*(A*c-c*C)+b*(c^2*C+A*d^2)+d*(A*b*c-b*c*C-a*A*d+a*C*d)*Tan[e+f*x]+b*C*(c^2+d^2)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[c^2+d^2,0] && LtQ[n,-1]
```

2: $\int(a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq0\,\wedge\,c^2+d^2\neq0\,\wedge\,n\nless-1$

Derivation: Algebraic expansion, nondegenerate tangent recurrence 1b with $c\to0,\ d\to1,\ A\to a\,c,\ B\to b\,c+a\,d,\ C\to b\,d,\ m\to1+m,\ n\to0,\ p\to0$ and algebraic simplification

Basis: $A+B\,z+C\,z^2==\frac{C\,(c+d\,z)^2}{d^2}-\frac{c^2\,C-A\,d^2+d\,(2\,c\,C-B\,d)\,z}{d^2}$

Rule: If $b\,c-a\,d\neq0\,\wedge\,c^2+d^2\neq0\,\wedge\,n\nless-1$, then

$$\int(a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx\ \longrightarrow$$

$$\frac{C}{d^2}\int(a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+2}\,dx-\frac{1}{d^2}\int(a+b\,\mathrm{Tan}[e+f\,x])\,(c+d\,\mathrm{Tan}[e+f\,x])^n\left(c^2\,C-A\,d^2+d\,(2\,c\,C-B\,d)\,\mathrm{Tan}[e+f\,x]\right)dx\ \longrightarrow$$

$$\frac{b\,C\,\mathrm{Tan}[e+f\,x]\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d\,f\,(n+2)}-\frac{1}{d\,(n+2)}\int(c+d\,\mathrm{Tan}[e+f\,x])^n\cdot$$
$$\left(b\,c\,C-a\,A\,d\,(n+2)-(A\,b+a\,B-b\,C)\,d\,(n+2)\,\mathrm{Tan}[e+f\,x]-(a\,C\,d\,(n+2)-b\,(c\,C-B\,d\,(n+2)))\,\mathrm{Tan}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])^n_.*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  b*C*Tan[e+f*x]*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+2)) -
  1/(d*(n+2))*Int[(c+d*Tan[e+f*x])^n*
     Simp[b*c*C-a*A*d*(n+2)-(A*b+a*B-b*C)*d*(n+2)*Tan[e+f*x]-(a*C*d*(n+2)-b*(c*C-B*d*(n+2)))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && NeQ[c^2+d^2,0] && Not[LtQ[n,-1]]
```

```
Int[(a_+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])^n_.*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  b*C*Tan[e+f*x]*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+2)) -
  1/(d*(n+2))*Int[(c+d*Tan[e+f*x])^n*
     Simp[b*c*C-a*A*d*(n+2)-(A*b-b*C)*d*(n+2)*Tan[e+f*x]-(a*C*d*(n+2)-b*c*C)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && NeQ[c^2+d^2,0] && Not[LtQ[n,-1]]
```

3. $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0$

1: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m<0$

Derivation: Algebraic expansion, singly degenerate tangent recurrence 2b with $A\to 1,\ B\to 0,\ p\to 0$ and algebraic simplification

Basis: If $a^2+b^2==0$, then $A+B\,z+C\,z^2==\frac{a\,A+b\,B-a\,C}{a}+\frac{(a+b\,z)\,(b\,B-a\,C+b\,C\,z)}{b^2}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m<0$, then

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx\,\longrightarrow$$

$$\frac{A\,b-a\,B-b\,C}{b}\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx+\frac{1}{b^2}\int(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(b\,B-a\,C+b\,C\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(a A + b B - a C)\,(a + b \,\mathrm{Tan}[e + f x])^m\,(c + d \,\mathrm{Tan}[e + f x])^{n+1}}{2 f m\,(b c - a d)} +$$

$$\frac{1}{2 a m\,(b c - a d)} \int (a + b \,\mathrm{Tan}[e + f x])^{m+1}\,(c + d \,\mathrm{Tan}[e + f x])^n \cdot$$

$$(b\,(c\,(A + C)\,m - B d\,(n + 1)) + a\,(B c m + C d\,(n + 1) - A d\,(2 m + n + 1)) + (b C d\,(m - n - 1) + A b d\,(m + n + 1) + a\,(2 c C m - B d\,(m + n + 1)))\,\mathrm{Tan}[e + f x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_.*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (a*A+b*B-a*C)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(2*f*m*(b*c-a*d)) +
  1/(2*a*m*(b*c-a*d))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
    Simp[b*(c*(A+C)*m-B*d*(n+1))+a*(B*c*m+C*d*(n+1)-A*d*(2*m+n+1))+
      (b*C*d*(m-n-1)+A*b*d*(m+n+1)+a*(2*c*C*m-B*d*(m+n+1)))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && (LtQ[m,0] || EqQ[m+n+1,0])
```

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_.*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  a*(A-C)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(2*f*m*(b*c-a*d)) +
  1/(2*a*m*(b*c-a*d))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
    Simp[b*c*(A+C)*m+a*(C*d*(n+1)-A*d*(2*m+n+1))+(b*C*d*(m-n-1)+A*b*d*(m+n+1)+2*a*c*C*m)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && (LtQ[m,0] || EqQ[m+n+1,0])
```

2. $\int (a + b \,\mathrm{Tan}[e + f x])^m\,(c + d \,\mathrm{Tan}[e + f x])^n\,(A + B \,\mathrm{Tan}[e + f x] + C \,\mathrm{Tan}[e + f x]^2)\,dx$ when $b c - a d \neq 0 \wedge a^2 + b^2 == 0 \wedge m \nless 0$

   1: $\int (a + b \,\mathrm{Tan}[e + f x])^m\,(c + d \,\mathrm{Tan}[e + f x])^n\,(A + B \,\mathrm{Tan}[e + f x] + C \,\mathrm{Tan}[e + f x]^2)\,dx$ when $b c - a d \neq 0 \wedge a^2 + b^2 == 0 \wedge m \nless 0 \wedge n < -1 \wedge c^2 + d^2 \neq 0$

Derivation: Algebraic expansion and singly degenerate tangent recurrence 1c with $A \to 1,\ B \to 0,\ p \to 0$

Basis: $A + B z + C z^2 == \frac{c^2 C - B c d + A d^2}{d^2} - \frac{(c + d z)\,(c C - B d - C d z)}{d^2}$

Rule: If $b c - a d \neq 0 \wedge a^2 + b^2 == 0 \wedge m \nless 0 \wedge n < -1 \wedge c^2 + d^2 \neq 0$, then

$$\int (a + b \,\mathrm{Tan}[e + f x])^m\,(c + d \,\mathrm{Tan}[e + f x])^n\,(A + B \,\mathrm{Tan}[e + f x] + C \,\mathrm{Tan}[e + f x]^2)\,dx \longrightarrow$$

$$\frac{c^2 C - B c d + A d^2}{d^2} \int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n \, dx - \frac{1}{d^2} \int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^{n+1} (c C - B d - C d \operatorname{Tan}[e + f x]) \, dx \longrightarrow$$

$$\frac{(c^2 C - B c d + A d^2) (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^{n+1}}{d f (n + 1) (c^2 + d^2)} - \frac{1}{a d (n + 1) (c^2 + d^2)} \int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^{n+1} \cdot$$
$$(b (c^2 C - B c d + A d^2) m - a d (n + 1) (A c - c C + B d) - a (d (B c - A d) (m + n + 1) - C (c^2 m - d^2 (n + 1))) \operatorname{Tan}[e + f x]) \, dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (c^2*C-B*c*d+A*d^2)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+1)*(c^2+d^2)) -
  1/(a*d*(n+1)*(c^2+d^2))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)*
    Simp[b*(c^2*C-B*c*d+A*d^2)*m-a*d*(n+1)*(A*c-c*C+B*d)-a*(d*(B*c-A*d)*(m+n+1)-C*(c^2*m-d^2*(n+1)))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && Not[LtQ[m,0]] && LtQ[n,-1] && NeQ[c^2+d^2,0]
```

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (c^2*C+A*d^2)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+1)*(c^2+d^2)) -
  1/(a*d*(n+1)*(c^2+d^2))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)*
    Simp[b*(c^2*C+A*d^2)*m-a*d*(n+1)*(A*c-c*C)-a*(-A*d^2*(m+n+1)-C*(c^2*m-d^2*(n+1)))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && Not[LtQ[m,0]] && LtQ[n,-1] && NeQ[c^2+d^2,0]
```

2: $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x] + C \operatorname{Tan}[e + f x]^2) \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 == 0 \land m \nless 0 \land m + n + 1 \neq 0$

Derivation: Algebraic expansion and singly degenerate tangent recurrence 2c with $A \to c,\ B \to d,\ n \to n + 1,\ p \to 0$

Basis: $A + B z + C z^2 == \frac{C (c+d z)^2}{d^2} + \frac{A d^2 - c^2 C - d (2 c C - B d) z}{d^2}$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 == 0 \land m \nless 0 \land m + n + 1 \neq 0$, then

$$\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n (A + B \operatorname{Tan}[e + f x] + C \operatorname{Tan}[e + f x]^2) \, dx \longrightarrow$$

$$\frac{C}{d^2}\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+2}\,dx+\frac{1}{d^2}\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A\,d^2-c^2\,C-d\,(2\,c\,C-B\,d)\,\mathrm{Tan}[e+f\,x]\right)dx\,\longrightarrow$$

$$\frac{C\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d\,f\,(m+n+1)}+$$

$$\frac{1}{b\,d\,(m+n+1)}\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A\,b\,d\,(m+n+1)+C\,(a\,c\,m-b\,d\,(n+1))-(C\,m\,(b\,c-a\,d)-b\,B\,d\,(m+n+1))\,\mathrm{Tan}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])^n_.*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  C*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n+1)) +
  1/(b*d*(m+n+1))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n*
    Simp[A*b*d*(m+n+1)+C*(a*c*m-b*d*(n+1))-(C*m*(b*c-a*d)-b*B*d*(m+n+1))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && Not[LtQ[m,0]] && NeQ[m+n+1,0]
```

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])^n_.*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  C*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n+1)) +
  1/(b*d*(m+n+1))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n*
    Simp[A*b*d*(m+n+1)+C*(a*c*m-b*d*(n+1))-C*m*(b*c-a*d)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && Not[LtQ[m,0]] && NeQ[m+n+1,0]
```

4. $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq0\,\land\,a^2+b^2\neq0\,\land\,c^2+d^2\neq0$

   1. $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq0\,\land\,a^2+b^2\neq0\,\land\,c^2+d^2\neq0\,\land\,m>0$

      1: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq0\,\land\,a^2+b^2\neq0\,\land\,c^2+d^2\neq0\,\land\,m>0\,\land\,n<-1$

Derivation: Nondegenerate tangent recurrence 1a with $p\to0$

Rule: If $b\,c-a\,d\neq0\,\land\,a^2+b^2\neq0\,\land\,c^2+d^2\neq0\,\land\,m>0\,\land\,n<-1$, then

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx\,\longrightarrow$$

$$\frac{\left(A\,d^2+c\,(c\,C-B\,d)\right)\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,\left(c^2+d^2\right)}-$$

$$\frac{1}{d\,(n+1)\,(c^2+d^2)}\int(a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}\cdot$$
$$\big(A\,d\,(b\,d\,m-a\,c\,(n+1))+(c\,C-B\,d)\,(b\,c\,m+a\,d\,(n+1))-$$
$$d\,(n+1)\,((A-C)\,(b\,c-a\,d)+B\,(a\,c+b\,d))\,\mathrm{Tan}[e+f\,x]-$$
$$b\,(d\,(B\,c-A\,d)\,(m+n+1)-C\,(c^2\,m-d^2\,(n+1)))\,\mathrm{Tan}[e+f\,x]^2\big)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (A*d^2+c*(c*C-B*d))*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+1)*(c^2+d^2)) -
  1/(d*(n+1)*(c^2+d^2))*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)*
    Simp[A*d*(b*d*m-a*c*(n+1))+(c*C-B*d)*(b*c*m+a*d*(n+1)) -
      d*(n+1)*((A-C)*(b*c-a*d)+B*(a*c+b*d))*Tan[e+f*x] -
      b*(d*(B*c-A*d)*(m+n+1)-C*(c^2*m-d^2*(n+1)))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,0] && LtQ[n,-1]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (A*d^2+c^2*C)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+1)*(c^2+d^2)) -
  1/(d*(n+1)*(c^2+d^2))*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)*
    Simp[A*d*(b*d*m-a*c*(n+1))+c*C*(b*c*m+a*d*(n+1)) -
      d*(n+1)*((A-C)*(b*c-a*d))*Tan[e+f*x] +
      b*(A*d^2*(m+n+1)+C*(c^2*m-d^2*(n+1)))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,0] && LtQ[n,-1]
```

2: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>0\,\wedge\,n\not<-1$

Derivation: Nondegenerate tangent recurrence 1b with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>0\,\wedge\,n\not<-1$, then

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx\longrightarrow$$
$$\frac{C\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d\,f\,(m+n+1)}+$$

$$\frac{1}{d\,(m+n+1)}\int(a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^n\cdot$$
$$(a\,A\,d\,(m+n+1)-C\,(b\,c\,m+a\,d\,(n+1))+d\,(A\,b+a\,B-b\,C)\,(m+n+1)\,\mathrm{Tan}[e+f\,x]-(C\,m\,(b\,c-a\,d)-b\,B\,d\,(m+n+1))\,\mathrm{Tan}[e+f\,x]^2)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  C*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n+1)) +
  1/(d*(m+n+1))*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^n*
    Simp[a*A*d*(m+n+1)-C*(b*c*m+a*d*(n+1))+d*(A*b+a*B-b*C)*(m+n+1)*Tan[e+f*x]-(C*m*(b*c-a*d)-b*B*d*(m+n+1))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,0] &&
  Not[IGtQ[n,0] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  C*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n+1)) +
  1/(d*(m+n+1))*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^n*
    Simp[a*A*d*(m+n+1)-C*(b*c*m+a*d*(n+1))+d*(A*b-b*C)*(m+n+1)*Tan[e+f*x]-C*m*(b*c-a*d)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,0] &&
  Not[IGtQ[n,0] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

2: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m<-1$

Derivation: Nondegenerate tangent recurrence 1c with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m<-1$, then

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2)\,dx\longrightarrow$$
$$\frac{(A\,b^2-a\,(b\,B-a\,C))\,(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{f\,(m+1)\,(b\,c-a\,d)\,(a^2+b^2)}+$$
$$\frac{1}{(m+1)\,(b\,c-a\,d)\,(a^2+b^2)}\int(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^n\cdot$$
$$(A\,(a\,(b\,c-a\,d)\,(m+1)-b^2\,d\,(m+n+2))+(b\,B-a\,C)\,(b\,c\,(m+1)+a\,d\,(n+1))-$$
$$(m+1)\,(b\,c-a\,d)\,(A\,b-a\,B-b\,C)\,\mathrm{Tan}[e+f\,x]-$$

$$d\left(A b^2 - a\left(b B - a C\right)\right)(m+n+2)\,\mathrm{Tan}[e+f x]^2\Big)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2-a*(b*B-a*C))*(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n+1)/(f*(m+1)*(b*c-a*d)*(a^2+b^2)) +
  1/((m+1)*(b*c-a*d)*(a^2+b^2))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
     Simp[A*(a*(b*c -a*d)*(m+1)-b^2*d*(m+n+2))+(b*B-a*C)*(b*c*(m+1)+a*d*(n+1)) -
       (m+1)*(b*c-a*d)*(A*b-a*B-b*C)*Tan[e+f*x] -
       d*(A*b^2-a*(b*B-a*C))*(m+n+2)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,-1] &&
  Not[ILtQ[n,-1] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2+a^2*C)*(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n+1)/(f*(m+1)*(b*c-a*d)*(a^2+b^2)) +
  1/((m+1)*(b*c-a*d)*(a^2+b^2))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
     Simp[A*(a*(b*c -a*d)*(m+1)-b^2*d*(m+n+2))-a*C*(b*c*(m+1)+a*d*(n+1)) -
       (m+1)*(b*c-a*d)*(A*b-b*C)*Tan[e+f*x] -
       d*(A*b^2+a^2*C)*(m+n+2)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,-1] &&
  Not[ILtQ[n,-1] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

3. $\int \frac{(c+d\,\mathrm{Tan}[e+f x])^n\,\left(A+B\,\mathrm{Tan}[e+f x]+C\,\mathrm{Tan}[e+f x]^2\right)}{a+b\,\mathrm{Tan}[e+f x]}\,dx$ when $b c-a d\neq 0\land a^2+b^2\neq 0\land c^2+d^2\neq 0\land n\not>0\land n\not<-1$

1: $\int \frac{A+B\,\mathrm{Tan}[e+f x]+C\,\mathrm{Tan}[e+f x]^2}{(a+b\,\mathrm{Tan}[e+f x])\,(c+d\,\mathrm{Tan}[e+f x])}\,dx$ when $b c-a d\neq 0\land a^2+b^2\neq 0\land c^2+d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{(a+b z)\,(c+d z)}==\frac{a\,(A c-c C+B d)+b\,(B c-A d+C d)}{\left(a^2+b^2\right)\left(c^2+d^2\right)}+\frac{\left(A b^2-a b B+a^2 C\right)(b-a z)}{(b c-a d)\left(a^2+b^2\right)(a+b z)}-\frac{\left(c^2 C-B c d+A d^2\right)(d-c z)}{(b c-a d)\left(c^2+d^2\right)(c+d z)}$

Rule: If $b c-a d\neq 0\land a^2+b^2\neq 0\land c^2+d^2\neq 0$, then

$$\int \frac{A + B \tan[e + f x] + C \tan[e + f x]^2}{(a + b \tan[e + f x]) (c + d \tan[e + f x])} dx \to$$
$$\frac{(a (A c - c C + B d) + b (B c - A d + C d)) x}{(a^2 + b^2) (c^2 + d^2)} + \frac{A b^2 - a b B + a^2 C}{(b c - a d) (a^2 + b^2)} \int \frac{b - a \tan[e + f x]}{a + b \tan[e + f x]} dx - \frac{c^2 C - B c d + A d^2}{(b c - a d) (c^2 + d^2)} \int \frac{d - c \tan[e + f x]}{c + d \tan[e + f x]} dx$$

Program code:

```
Int[(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2)/((a_+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])),x_Symbol] :=
  (a*(A*c-c*C+B*d)+b*(B*c-A*d+C*d))*x/((a^2+b^2)*(c^2+d^2)) +
  (A*b^2-a*b*B+a^2*C)/((b*c-a*d)*(a^2+b^2))*Int[(b-a*Tan[e+f*x])/(a+b*Tan[e+f*x]),x] -
  (c^2*C-B*c*d+A*d^2)/((b*c-a*d)*(c^2+d^2))*Int[(d-c*Tan[e+f*x])/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

```
Int[(A_.+C_.*tan[e_.+f_.*x_]^2)/((a_+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])),x_Symbol] :=
  (a*(A*c-c*C)-b*(A*d-C*d))*x/((a^2+b^2)*(c^2+d^2)) +
  (A*b^2+a^2*C)/((b*c-a*d)*(a^2+b^2))*Int[(b-a*Tan[e+f*x])/(a+b*Tan[e+f*x]),x] -
  (c^2*C+A*d^2)/((b*c-a*d)*(c^2+d^2))*Int[(d-c*Tan[e+f*x])/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

2: $\int \frac{(c + d \tan[e + f x])^n (A + B \tan[e + f x] + C \tan[e + f x]^2)}{a + b \tan[e + f x]} dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land n \not> 0 \land n \not\leq -1$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{a+b z} == \frac{b B+a (A-C)-(A b-a B-b C) z}{a^2+b^2} + \frac{(A b^2-a b B+a^2 C) (1+z^2)}{(a^2+b^2) (a+b z)}$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land n \not> 0 \land n \not\leq -1$, then

$$\int \frac{(c + d \tan[e + f x])^n (A + B \tan[e + f x] + C \tan[e + f x]^2)}{a + b \tan[e + f x]} dx \to$$
$$\frac{1}{a^2 + b^2} \int (c + d \tan[e + f x])^n (b B + a (A - C) + (a B - b (A - C)) \tan[e + f x]) dx +$$

$$\frac{A\,b^2-a\,b\,B+a^2\,C}{a^2+b^2}\int\frac{(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(1+\mathrm{Tan}[e+f\,x]^2\right)}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2)/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(a^2+b^2)*Int[(c+d*Tan[e+f*x])^n*Simp[b*B+a*(A-C)+(a*B-b*(A-C))*Tan[e+f*x],x],x] +
  (A*b^2-a*b*B+a^2*C)/(a^2+b^2)*Int[(c+d*Tan[e+f*x])^n*(1+Tan[e+f*x]^2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && Not[GtQ[n,0]] && Not[LeQ[n,-1]]
```

```
Int[(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+C_.*tan[e_.+f_.*x_]^2)/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(a^2+b^2)*Int[(c+d*Tan[e+f*x])^n*Simp[a*(A-C)-(A*b-b*C)*Tan[e+f*x],x],x] +
  (A*b^2+a^2*C)/(a^2+b^2)*Int[(c+d*Tan[e+f*x])^n*(1+Tan[e+f*x]^2)/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && Not[GtQ[n,0]] && Not[LeQ[n,-1]]
```

4: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0$

Derivation: Integration by substitution

Basis: $F[\mathrm{Tan}[e+f\,x]] == \frac{1}{f}\,\mathrm{Subst}\left[\frac{F[x]}{1+x^2},\,x,\,\mathrm{Tan}[e+f\,x]\right]\partial_x\mathrm{Tan}[e+f\,x]$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0$, then

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\left(A+B\,\mathrm{Tan}[e+f\,x]+C\,\mathrm{Tan}[e+f\,x]^2\right)dx\,\longrightarrow$$

$$\frac{1}{f}\,\mathrm{Subst}\left[\int\frac{(a+b\,x)^m\,(c+d\,x)^n\,\left(A+B\,x+C\,x^2\right)}{1+x^2}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+B_.*tan[e_.+f_.*x_]+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+b*ff*x)^m*(c+d*ff*x)^n*(A+B*ff*x+C*ff^2*x^2)/(1+ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_*(A_.+C_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+b*ff*x)^m*(c+d*ff*x)^n*(A+C*ff^2*x^2)/(1+ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```