Rules for integrands of the form $\left(a + b x^2 + c x^4\right)^p$

1. $\int \left(a + b x^2 + c x^4\right)^p \, dx$ when $b^2 - 4 a c == 0$

x: $\int \left(a + b x^2 + c x^4\right)^p \, dx$ when $b^2 - 4 a c == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2 - 4 a c == 0$, then $a + b z + c z^2 == \frac{1}{c} \left(\frac{b}{2} + c z\right)^2$

Rule 1.2.2.1.1.1: If $b^2 - 4 a c == 0 \land p \in \mathbb{Z}$, then

$$\int \left(a + b x^2 + c x^4\right)^p \, dx \longrightarrow \frac{1}{c^p} \int \left(\frac{b}{2} + c x^2\right)^{2 p} \, dx$$

Program code:

```
(* Int[(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  1/c^p*Int[(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

2. $\int (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c == 0 \land p \notin \mathbb{Z}$

x: $\int \frac{1}{(a + b x^2 + c x^4)^{5/4}} \, dx$ when $b^2 - 4 a c == 0$

Derivation: Square trinomial recurrence 2c with $m + 4 (p + 1) + 1 == 0$

Rule 1.2.2.1.1.2.1: If $b^2 - 4 a c == 0$, then

$$\int \frac{1}{(a + b x^2 + c x^4)^{5/4}} \, dx \longrightarrow \frac{2 x}{3 a (a + b x^2 + c x^4)^{1/4}} + \frac{x (2 a + b x^2)}{6 a (a + b x^2 + c x^4)^{5/4}}$$

Program code:

```
(* Int[1/(a_+b_.*x_^2+c_.*x_^4)^(5/4),x_Symbol] :=
  2*x/(3*a*(a+b*x^2+c*x^4)^(1/4)) + x*(2*a+b*x^2)/(6*a*(a+b*x^2+c*x^4)^(5/4)) /;
FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0] *)
```

2: $\int \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4 a c == 0$, then $\partial_x \frac{\left(a+b x^2+c x^4\right)^p}{\left(b+2 c x^2\right)^{2 p}} == 0$

Note: If $b^2-4 a c == 0$, then $a+b z+c z^2 == \frac{1}{4 c} (b+2 c z)^2$

Rule 1.2.2.1.1.2.2: If $b^2-4 a c == 0 \land p \notin \mathbb{Z}$, then

$$\int \left(a+b x^2+c x^4\right)^p dx \longrightarrow \frac{\left(a+b x^2+c x^4\right)^p}{\left(b+2 c x^2\right)^{2 p}} \int \left(b+2 c x^2\right)^{2 p} dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^p/(b+2*c*x^2)^(2*p)*Int[(b+2*c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2]
```

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/(1+2*c*x^2/b)^(2*FracPart[p])*Int[(1+2*c*x^2/b)^(2*p),x] /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```

2. $\int (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c \neq 0 \land p > 0$

1: $\int (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c \neq 0 \land p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.2.1.2.1: If $b^2 - 4 a c \neq 0 \land p \in \mathbb{Z}^+$, then

$$\int (a + b x^2 + c x^4)^p \, dx \rightarrow \int \text{ExpandIntegrand}\left[(a + b x^2 + c x^4)^p, x\right] dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0]
```

2: $\int (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c \neq 0 \land p > 0$

Derivation: Trinomial recurrence 1b with $m = 0$, $A = 1$ and $B = 0$

Rule 1.2.2.1.2.2: If $b^2 - 4 a c \neq 0 \land p > 0$, then

$$\int (a + b x^2 + c x^4)^p \, dx \rightarrow \frac{x (a + b x^2 + c x^4)^p}{4 p + 1} + \frac{2 p}{4 p + 1} \int (2 a + b x^2) (a + b x^2 + c x^4)^{p-1} \, dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  x*(a+b*x^2+c*x^4)^p/(4*p+1) +
  2*p/(4*p+1)*Int[(2*a+b*x^2)*(a+b*x^2+c*x^4)^(p-1),x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && IntegerQ[2*p]
```

3: $\int \left(a + b x^2 + c x^4\right)^p \, dx$ when $b^2 - 4 a c \neq 0 \land p < -1$

Reference: G&R 2.161.5

Derivation: Trinomial recurrence 2b with $m = 0$, $A = 1$ and $B = 0$

Note: G&R 2.161.4 is a special case of G&R 2.161.5.

Rule 1.2.2.1.3: If $b^2 - 4 a c \neq 0 \land p < -1$, then

$$\int \left(a + b x^2 + c x^4\right)^p \, dx \longrightarrow$$

$$-\frac{x \left(b^2 - 2 a c + b c x^2\right) \left(a + b x^2 + c x^4\right)^{p+1}}{2 a (p + 1) \left(b^2 - 4 a c\right)} +$$

$$\frac{1}{2 a (p + 1) \left(b^2 - 4 a c\right)} \int \left(b^2 - 2 a c + 2 (p + 1) \left(b^2 - 4 a c\right) + b c (4 p + 7) x^2\right) \left(a + b x^2 + c x^4\right)^{p+1} \, dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  -x*(b^2-2*a*c+b*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*a*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*Int[(b^2-2*a*c+2*(p+1)*(b^2-4*a*c)+b*c*(4*p+7)*x^2)*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IntegerQ[2*p]
```

4. $\int \frac{1}{a + b x^2 + c x^4} \, dx$ when $b^2 - 4 a c \neq 0$

1: $\int \frac{1}{a + b x^2 + c x^4} \, dx$ when $b^2 - 4 a c \neq 0 \land b^2 - 4 a c > 0$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let $q \to \sqrt{b^2 - 4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2} == \frac{c}{q}\,\frac{1}{\frac{b}{2}-\frac{q}{2}+c\,z} - \frac{c}{q}\,\frac{1}{\frac{b}{2}+\frac{q}{2}+c\,z}$

- Rule 1.2.2.1.4.1: If $b^2 - 4\,a\,c \neq 0$, let $q \to \sqrt{b^2 - 4\,a\,c}$, then

$$\int \frac{1}{a+b\,x^2+c\,x^4}\,dx \longrightarrow \frac{c}{q}\int \frac{1}{\frac{b}{2}-\frac{q}{2}+c\,x^2}\,dx - \frac{c}{q}\int \frac{1}{\frac{b}{2}+\frac{q}{2}+c\,x^2}\,dx$$

Program code:

```
Int[1/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  c/q*Int[1/(b/2-q/2+c*x^2),x] - c/q*Int[1/(b/2+q/2+c*x^2),x]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && PosQ[b^2-4*a*c]
```

2: $\int \frac{1}{a+b x^2+c x^4}\,dx$ when $b^2-4 a c \neq 0 \wedge b^2-4 a c \not> 0$

Derivation: Algebraic expansion

Basis: If $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2 q-\frac{b}{c}}$, then $\frac{1}{a+b z^2+c z^4} == \frac{r-z}{2 c q r\,(q-r z+z^2)}+\frac{r+z}{2 c q r\,(q+r z+z^2)}$

Note: If $(a \mid b \mid c) \in \mathbb{R} \wedge b^2-4 a c<0$, then $\frac{a}{c}>0$ and $2\sqrt{\frac{a}{c}}-\frac{b}{c}>0$.

Rule 1.2.2.1.4.2: If $b^2-4 a c \neq 0 \wedge b^2-4 a c \not> 0$, let $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2 q-\frac{b}{c}}$, then

$$\int \frac{1}{a+b x^2+c x^4}\,dx \longrightarrow \frac{1}{2 c q r}\int \frac{r-x}{q-r x+x^2}\,dx+\frac{1}{2 c q r}\int \frac{r+x}{q+r x+x^2}\,dx$$

Program code:

```
Int[1/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*q*r)*Int[(r-x)/(q-r*x+x^2),x] + 1/(2*c*q*r)*Int[(r+x)/(q+r*x+x^2),x]]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && NegQ[b^2-4*a*c]
```

5. $\int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c \neq 0$

1. $\int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c>0$

1: $\int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c>0 \wedge c<0$

Derivation: Algebraic expansion

Basis: If $b^2-4ac>0 \land c<0$, let $q \to \sqrt{b^2-4ac}$, then

$$\sqrt{a+bx^2+cx^4} = \frac{1}{2\sqrt{-c}}\sqrt{b+q+2cx^2}\sqrt{-b+q-2cx^2}$$

- Rule 1.2.2.1.5.1.1: If $b^2-4ac>0 \land c<0$, let $q \to \sqrt{b^2-4ac}$, then

$$\int \frac{1}{\sqrt{a+bx^2+cx^4}}\,dx \longrightarrow 2\sqrt{-c}\int \frac{1}{\sqrt{b+q+2cx^2}\sqrt{-b+q-2cx^2}}\,dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*Sqrt[-c]*Int[1/(Sqrt[b+q+2*c*x^2]*Sqrt[-b+q-2*c*x^2]),x]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[c,0]
```

2. $\int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land c \not< 0$

1: $\int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{c}{a}>0 \land \frac{b}{a}<0$

Reference: G&R 3.165.2

Derivation: Piecewise constant extraction

Basis: Let $q=\left(\frac{c}{a}\right)^{1/4}$, then $\partial_x \frac{\left(1+q^2 x^2\right) \sqrt{\frac{\left(a+b x^2+c x^4\right)}{a \left(1+q^2 x^2\right)^2}}}{\sqrt{a+b x^2+c x^4}}==0$

Rule 1.2.2.1.5.1.2.1: If $b^2-4 a c>0 \land \frac{c}{a}>0 \land \frac{b}{a}<0$, let $q \to \left(\frac{c}{a}\right)^{1/4}$, then

$$\int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx \to \frac{\left(1+q^2 x^2\right) \sqrt{\frac{\left(a+b x^2+c x^4\right)}{a \left(1+q^2 x^2\right)^2}}}{2 q \sqrt{a+b x^2+c x^4}} \text{EllipticF}\left[2 \text{ArcTan}[q x], \frac{1}{2}-\frac{b q^2}{4 c}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,4]},
  (1+q^2*x^2)*Sqrt[(a+b*x^2+c*x^4)/(a*(1+q^2*x^2)^2)]/(2*q*Sqrt[a+b*x^2+c*x^4])*EllipticF[2*ArcTan[q*x],1/2-b*q^2/(4*c)]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && GtQ[c/a,0] && LtQ[b/a,0]
```

2: $\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c>0\,\wedge\,a<0\,\wedge\,c>0$

Reference: G&R 3.152.3+

Note: Not sure if the shorter rule is valid for all q.

- Rule 1.2.2.1.5.1.2.2: If $b^2-4\,a\,c>0\,\wedge\,a<0\,\wedge\,c>0$, let $q\to\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx \to \frac{\sqrt{\frac{2\,a+(b-q)\,x^2}{2\,a+(b+q)\,x^2}}\,\sqrt{\frac{2\,a+(b+q)\,x^2}{q}}}{2\,\sqrt{a+b\,x^2+c\,x^4}\,\sqrt{\frac{a}{2\,a+(b+q)\,x^2}}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{x}{\sqrt{\frac{2\,a+(b+q)\,x^2}{2\,q}}}\right],\frac{b+q}{2\,q}\right]$$

$$\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx \to \frac{\sqrt{-2\,a-(b-q)\,x^2}\,\sqrt{\frac{2\,a+(b+q)\,x^2}{q}}}{2\,\sqrt{-a}\,\sqrt{a+b\,x^2+c\,x^4}}\,\text{EllipticF}\left[\text{ArcSin}\left[\frac{x}{\sqrt{\frac{2\,a+(b+q)\,x^2}{2\,q}}}\right],\frac{b+q}{2\,q}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[-2*a-(b-q)*x^2]*Sqrt[(2*a+(b+q)*x^2)/q]/(2*Sqrt[-a]*Sqrt[a+b*x^2+c*x^4])*
    EllipticF[ArcSin[x/Sqrt[(2*a+(b+q)*x^2)/(2*q)]],(b+q)/(2*q)] /;
  IntegerQ[q]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[a,0] && GtQ[c,0]
```

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[(2*a+(b-q)*x^2)/(2*a+(b+q)*x^2)]*Sqrt[(2*a+(b+q)*x^2)/q]/(2*Sqrt[a+b*x^2+c*x^4]*Sqrt[a/(2*a+(b+q)*x^2)])*
    EllipticF[ArcSin[x/Sqrt[(2*a+(b+q)*x^2)/(2*q)]],(b+q)/(2*q)]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[a,0] && GtQ[c,0]
```

3. $\int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{b\pm\sqrt{b^2-4 a c}}{a}>0$

1: $\int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{b+\sqrt{b^2-4 a c}}{a}>0$

Reference: G&R 3.152.1+

- Rule 1.2.2.1.5.1.2.3.1: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b+q}{a}>0$, then

$$\int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx \to \frac{\left(2 a+(b+q) x^2\right) \sqrt{\frac{2 a+(b-q) x^2}{2 a+(b+q) x^2}}}{2 a \sqrt{\frac{b+q}{2 a}} \sqrt{a+b x^2+c x^4}} \, \text{EllipticF}\left[\text{ArcTan}\left[\sqrt{\frac{b+q}{2 a}} \, x\right], \frac{2 q}{b+q}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (2*a+(b+q)*x^2)*Sqrt[(2*a+(b-q)*x^2)/(2*a+(b+q)*x^2)]/(2*a*Rt[(b+q)/(2*a),2]*Sqrt[a+b*x^2+c*x^4])*
    EllipticF[ArcTan[Rt[(b+q)/(2*a),2]*x],2*q/(b+q)] /;
 PosQ[(b+q)/a] && Not[PosQ[(b-q)/a] && SimplerSqrtQ[(b-q)/(2*a),(b+q)/(2*a)]]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2: $\int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{b-\sqrt{b^2-4 a c}}{a}>0$

Reference: G&R 3.152.1-

- Rule 1.2.2.1.5.1.2.3.2: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b-q}{a}>0$ then

$$\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx \longrightarrow \frac{\left(2\,a+(b-q)\,x^2\right)\sqrt{\frac{2\,a+(b+q)\,x^2}{2\,a+(b-q)\,x^2}}}{2\,a\,\sqrt{\frac{b-q}{2\,a}}\,\sqrt{a+b\,x^2+c\,x^4}}\,\text{EllipticF}\left[\text{ArcTan}\left[\sqrt{\frac{b-q}{2\,a}}\,x\right],-\frac{2\,q}{b-q}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (2*a+(b-q)*x^2)*Sqrt[(2*a+(b+q)*x^2)/(2*a+(b-q)*x^2)]/(2*a*Rt[(b-q)/(2*a),2]*Sqrt[a+b*x^2+c*x^4])*
    EllipticF[ArcTan[Rt[(b-q)/(2*a),2]*x],-2*q/(b-q)] /;
 PosQ[(b-q)/a]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

4. $\int \frac{1}{\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c>0 \land \frac{b\pm\sqrt{b^2-4 a c}}{a} \not> 0$

1: $\int \frac{1}{\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c>0 \land \frac{b+\sqrt{b^2-4 a c}}{a} \not> 0$

Reference: G&R 3.152.7+

Rule 1.2.2.1.5.1.2.4.1: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b+q}{a} \not> 0$ then

$$\int \frac{1}{\sqrt{a+b x^2+c x^4}} dx \to \frac{\sqrt{1+\frac{(b+q) x^2}{2 a}} \sqrt{1+\frac{(b-q) x^2}{2 a}}}{\sqrt{-\frac{b+q}{2 a}} \sqrt{a+b x^2+c x^4}} \text{EllipticF}\left[\text{ArcSin}\left[\sqrt{-\frac{b+q}{2 a}} x\right], \frac{b-q}{b+q}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+(b+q)*x^2/(2*a)]*Sqrt[1+(b-q)*x^2/(2*a)]/(Rt[-(b+q)/(2*a),2]*Sqrt[a+b*x^2+c*x^4])*
    EllipticF[ArcSin[Rt[-(b+q)/(2*a),2]*x],(b-q)/(b+q)] /;
 NegQ[(b+q)/a] && Not[NegQ[(b-q)/a] && SimplerSqrtQ[-(b-q)/(2*a),-(b+q)/(2*a)]]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2: $\int \frac{1}{\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c>0 \land \frac{b-\sqrt{b^2-4 a c}}{a} \not> 0$

Reference: G&R 3.152.7-

Rule 1.2.2.1.5.1.2.4.2: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b-q}{a} \not> 0$ then

$$\int \frac{1}{\sqrt{a + b\,x^2 + c\,x^4}}\,dx \,\longrightarrow\, \frac{\sqrt{1 + \frac{(b-q)\,x^2}{2\,a}}\,\sqrt{1 + \frac{(b+q)\,x^2}{2\,a}}}{\sqrt{-\frac{b-q}{2\,a}}\,\sqrt{a + b\,x^2 + c\,x^4}}\,\text{EllipticF}\left[\text{ArcSin}\left[\sqrt{-\frac{b-q}{2\,a}}\,x\right],\,\frac{b+q}{b-q}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+(b-q)*x^2/(2*a)]*Sqrt[1+(b+q)*x^2/(2*a)]/(Rt[-(b-q)/(2*a),2]*Sqrt[a+b*x^2+c*x^4])*
    EllipticF[ArcSin[Rt[-(b-q)/(2*a),2]*x],(b+q)/(b-q)] /;
 NegQ[(b-q)/a]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2. $\int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \not> 0$

1: $\int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land \frac{c}{a} > 0$

Reference: G&R 3.165.2

Derivation: Piecewise constant extraction

Basis: Let $q = \left(\frac{c}{a}\right)^{1/4}$, then $\partial_x \frac{\left(1+q^2 x^2\right) \sqrt{\frac{\left(a+b x^2+c x^4\right)}{a \left(1+q^2 x^2\right)^2}}}{\sqrt{a+b x^2+c x^4}} == 0$

Rule 1.2.2.1.5.2.1: If $b^2 - 4 a c \neq 0 \land \frac{c}{a} > 0$, let $q \to \left(\frac{c}{a}\right)^{1/4}$, then

$$\int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx \to \frac{\left(1 + q^2 x^2\right) \sqrt{\frac{\left(a+b x^2+c x^4\right)}{a \left(1+q^2 x^2\right)^2}}}{2 q \sqrt{a + b x^2 + c x^4}} \text{EllipticF}\left[2 \text{ArcTan}[q x], \frac{1}{2} - \frac{b q^2}{4 c}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,4]},
  (1+q^2*x^2)*Sqrt[(a+b*x^2+c*x^4)/(a*(1+q^2*x^2)^2)]/(2*q*Sqrt[a+b*x^2+c*x^4])*EllipticF[2*ArcTan[q*x],1/2-b*q^2/(4*c)]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a]
```

2: $\int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land \frac{c}{a} \not> 0$

Derivation: Piecewise constant extraction

Basis: If $q \to \sqrt{b^2-4\,a\,c}$, then $\partial_x \frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}} = 0$

- Rule 1.2.2.1.5.2.2: If $b^2-4\,a\,c \neq 0 \land \frac{c}{a} \not> 0$, let $q \to \sqrt{b^2-4\,a\,c}$, then

$$\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx \to \frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}} \int \frac{1}{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}\,dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
    Int[1/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && NegQ[c/a]
```

6: $\int (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c \neq 0$

Derivation: Piecewise constant extraction

Basis: If $q \to \sqrt{b^2 - 4 a c}$, then $\partial_x \frac{(a+b x^2+c x^4)^p}{\left(1+\frac{2 c x^2}{b+q}\right)^p \left(1+\frac{2 c x^2}{b-q}\right)^p} == 0$

Rule 1.2.2.1.6: If $b^2 - 4 a c \neq 0$, let $q \to \sqrt{b^2 - 4 a c}$, then

$$\int (a + b x^2 + c x^4)^p \, dx \longrightarrow \frac{a^{\text{IntPart}[p]} (a + b x^2 + c x^4)^{\text{FracPart}[p]}}{\left(1 + \frac{2 c x^2}{b+q}\right)^{\text{FracPart}[p]} \left(1 + \frac{2 c x^2}{b-q}\right)^{\text{FracPart}[p]}} \int \left(1 + \frac{2 c x^2}{b + q}\right)^p \left(1 + \frac{2 c x^2}{b - q}\right)^p dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/((1+2*c*x^2/(b+q))^FracPart[p]*(1+2*c*x^2/(b-q))^FracPart[p])*
    Int[(1+2*c*x^2/(b+q))^p*(1+2*c*x^2/(b-q))^p,x]] /;
FreeQ[{a,b,c,p},x] && NeQ[b^2-4*a*c,0]
```

S: $\int (a + b x + c x^2 + d x^3 + e x^4)^p \, dx$ when $d^3 - 4 c d e + 8 b e^2 == 0 \land p \notin \{1, 2, 3\}$

Derivation: Integration by substitution

Basis: If $d^3 - 4 c d e + 8 b e^2 == 0$, then
$(a + b x + c x^2 + d x^3 + e x^4)^p == \text{Subst}\left[\left(a + \frac{d^4}{256 e^3} - \frac{b d}{8 e} + \left(c - \frac{3 d^2}{8 e}\right) x^2 + e x^4\right)^p, x, \frac{d}{4 e} + x\right] \partial_x \left(\frac{d}{4 e} + x\right)$

Note: The substitution transforms a dense quartic polynomial into a symmetric quartic trinomial.

Rule: If $d^3 - 4 c d e + 8 b e^2 == 0 \land p \notin \{1, 2, 3\}$, then

$$\int \left(a + b\,x + c\,x^2 + d\,x^3 + e\,x^4\right)^p\,dx \;\longrightarrow\; \text{Subst}\left[\int \left(a + \frac{d^4}{256\,e^3} - \frac{b\,d}{8\,e} + \left(c - \frac{3\,d^2}{8\,e}\right)x^2 + e\,x^4\right)^p dx,\; x,\; \frac{d}{4\,e} + x\right]$$

Program code:

```
Int[P4_^p_,x_Symbol] :=
  With[{a=Coeff[P4,x,0],b=Coeff[P4,x,1],c=Coeff[P4,x,2],d=Coeff[P4,x,3],e=Coeff[P4,x,4]},
  Subst[Int[SimplifyIntegrand[(a+d^4/(256*e^3)-b*d/(8*e)+(c-3*d^2/(8*e))*x^2+e*x^4)^p,x],x],x,d/(4*e)+x] /;
 EqQ[d^3-4*c*d*e+8*b*e^2,0] && NeQ[d,0]] /;
FreeQ[p,x] && PolyQ[P4,x,4] && NeQ[p,2] && NeQ[p,3]
```