Rules for integrands of the form $F^{c\,(a+b\,x)}\,\text{Hyper}[d+e\,x]^n$

1. $\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx$

    1. $\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx$ when $e^2\,n^2-b^2\,c^2\,\text{Log}[F]^2\neq 0\wedge n>0$

        1: $\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]\,dx$ when $e^2-b^2\,c^2\,\text{Log}[F]^2\neq 0$

Reference: CRC 533h

Reference: CRC 538h

Rule: If $e^2-b^2\,c^2\,\text{Log}[F]^2\neq 0$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]\,dx\;\longrightarrow\;-\frac{b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]}{e^2-b^2\,c^2\,\text{Log}[F]^2}+\frac{e\,F^{c\,(a+b\,x)}\,\text{Cosh}[d+e\,x]}{e^2-b^2\,c^2\,\text{Log}[F]^2}$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_],x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sinh[d+e*x]/(e^2-b^2*c^2*Log[F]^2) +
  e*F^(c*(a+b*x))*Cosh[d+e*x]/(e^2-b^2*c^2*Log[F]^2) /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2-b^2*c^2*Log[F]^2,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_],x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Cosh[d+e*x]/(e^2-b^2*c^2*Log[F]^2) +
  e*F^(c*(a+b*x))*Sinh[d+e*x]/(e^2-b^2*c^2*Log[F]^2) /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2-b^2*c^2*Log[F]^2,0]
```

2: $\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx$ when $e^2\,n^2-b^2\,c^2\,\text{Log}[F]^2\neq 0\,\wedge\,n>1$

Reference: CRC 542h

Reference: CRC 543h

Rule: If $e^2\,n^2-b^2\,c^2\,\text{Log}[F]^2\neq 0\,\wedge\,n>1$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx\,\longrightarrow$$
$$-\frac{b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n}{e^2\,n^2-b^2\,c^2\,\text{Log}[F]^2}+\frac{e\,n\,F^{c\,(a+b\,x)}\,\text{Cosh}[d+e\,x]\,\text{Sinh}[d+e\,x]^{n-1}}{e^2\,n^2-b^2\,c^2\,\text{Log}[F]^2}-\frac{n\,(n-1)\,e^2}{e^2\,n^2-b^2\,c^2\,\text{Log}[F]^2}\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^{n-2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sinh[d+e*x]^n/(e^2*n^2-b^2*c^2*Log[F]^2) +
  e*n*F^(c*(a+b*x))*Cosh[d+e*x]*Sinh[d+e*x]^(n-1)/(e^2*n^2-b^2*c^2*Log[F]^2) -
  n*(n-1)*e^2/(e^2*n^2-b^2*c^2*Log[F]^2)*Int[F^(c*(a+b*x))*Sinh[d+e*x]^(n-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*n^2-b^2*c^2*Log[F]^2,0] && GtQ[n,1]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Cosh[d+e*x]^n/(e^2*n^2-b^2*c^2*Log[F]^2) +
  e*n*F^(c*(a+b*x))*Sinh[d+e*x]*Cosh[d+e*x]^(n-1)/(e^2*n^2-b^2*c^2*Log[F]^2) +
  n*(n-1)*e^2/(e^2*n^2-b^2*c^2*Log[F]^2)*Int[F^(c*(a+b*x))*Cosh[d+e*x]^(n-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*n^2-b^2*c^2*Log[F]^2,0] && GtQ[n,1]
```

2: $\int F^{c (a+b x)} \operatorname{Sinh}[d+e x]^n \, dx$ when $e^2 (n+2)^2 - b^2 c^2 \operatorname{Log}[F]^2 == 0 \land n \neq -1 \land n \neq -2$

Reference: CRC 551h when $e^2 (n+2)^2 - b^2 c^2 \operatorname{Log}[F]^2 == 0$

Reference: CRC 552h when $e^2 (n+2)^2 - b^2 c^2 \operatorname{Log}[F]^2 == 0$

Rule: If $e^2 (n+2)^2 - b^2 c^2 \operatorname{Log}[F]^2 == 0 \land n \neq -1 \land n \neq -2$, then

$$\int F^{c (a+b x)} \operatorname{Sinh}[d+e x]^n \, dx \longrightarrow -\frac{b c \operatorname{Log}[F] F^{c (a+b x)} \operatorname{Sinh}[d+e x]^{n+2}}{e^2 (n+1) (n+2)} + \frac{F^{c (a+b x)} \operatorname{Cosh}[d+e x] \operatorname{Sinh}[d+e x]^{n+1}}{e (n+1)}$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sinh[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) +
  F^(c*(a+b*x))*Cosh[d+e*x]*Sinh[d+e*x]^(n+1)/(e*(n+1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[e^2*(n+2)^2-b^2*c^2*Log[F]^2,0] && NeQ[n,-1] && NeQ[n,-2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_]^n_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Cosh[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) -
  F^(c*(a+b*x))*Sinh[d+e*x]*Cosh[d+e*x]^(n+1)/(e*(n+1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[e^2*(n+2)^2-b^2*c^2*Log[F]^2,0] && NeQ[n,-1] && NeQ[n,-2]
```

3: $\int F^{c (a+b x)} \operatorname{Sinh}[d+e x]^n \, dx$ when $e^2 (n+2)^2 - b^2 c^2 \operatorname{Log}[F]^2 \neq 0 \land n < -1 \land n \neq -2$

Reference: CRC 551h, CRC 542h inverted

Reference: CRC 552h, CRC 543h inverted

Rule: If $e^2 (n+2)^2 - b^2 c^2 \operatorname{Log}[F]^2 \neq 0 \land n < -1 \land n \neq -2$, then

$$\int F^{c (a+b x)} \operatorname{Sinh}[d+e x]^n \, dx \longrightarrow$$

$$-\frac{b\,c\,\mathrm{Log}[F]\,F^{c\,(a+b\,x)}\,\mathrm{Sinh}[d+e\,x]^{n+2}}{e^2\,(n+1)\,(n+2)}+\frac{F^{c\,(a+b\,x)}\,\mathrm{Cosh}[d+e\,x]\,\mathrm{Sinh}[d+e\,x]^{n+1}}{e\,(n+1)}-\frac{e^2\,(n+2)^2-b^2\,c^2\,\mathrm{Log}[F]^2}{e^2\,(n+1)\,(n+2)}\int F^{c\,(a+b\,x)}\,\mathrm{Sinh}[d+e\,x]^{n+2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sinh[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) +
  F^(c*(a+b*x))*Cosh[d+e*x]*Sinh[d+e*x]^(n+1)/(e*(n+1)) -
  (e^2*(n+2)^2-b^2*c^2*Log[F]^2)/(e^2*(n+1)*(n+2))*Int[F^(c*(a+b*x))*Sinh[d+e*x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*(n+2)^2-b^2*c^2*Log[F]^2,0] && LtQ[n,-1] && NeQ[n,-2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_]^n_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Cosh[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) -
  F^(c*(a+b*x))*Sinh[d+e*x]*Cosh[d+e*x]^(n+1)/(e*(n+1)) +
  (e^2*(n+2)^2-b^2*c^2*Log[F]^2)/(e^2*(n+1)*(n+2))*Int[F^(c*(a+b*x))*Cosh[d+e*x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*(n+2)^2-b^2*c^2*Log[F]^2,0] && LtQ[n,-1] && NeQ[n,-2]
```

4: $\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\text{Sinh}[z]=\frac{1}{2}\,e^{-z}\left(-1+e^{2\,z}\right)$

Basis: $\partial_x\frac{e^{n\,(d+e\,x)}\,\text{Sinh}[d+e\,x]^n}{\left(-1+e^{2\,(d+e\,x)}\right)^n}=0$

Rule: If $n\notin\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx\;\longrightarrow\;\frac{e^{n\,(d+e\,x)}\,\text{Sinh}[d+e\,x]^n}{\left(-1+e^{2\,(d+e\,x)}\right)^n}\int F^{c\,(a+b\,x)}\,\frac{\left(-1+e^{2\,(d+e\,x)}\right)^n}{e^{n\,(d+e\,x)}}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^n_,x_Symbol] :=
  E^(n*(d+e*x))*Sinh[d+e*x]^n/(-1+E^(2*(d+e*x)))^n*Int[F^(c*(a+b*x))*(-1+E^(2*(d+e*x)))^n/E^(n*(d+e*x)),x] /;
FreeQ[{F,a,b,c,d,e,n},x] && Not[IntegerQ[n]]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_]^n_,x_Symbol] :=
  E^(n*(d+e*x))*Cosh[d+e*x]^n/(1+E^(2*(d+e*x)))^n*Int[F^(c*(a+b*x))*(1+E^(2*(d+e*x)))^n/E^(n*(d+e*x)),x] /;
FreeQ[{F,a,b,c,d,e,n},x] && Not[IntegerQ[n]]
```

2: $\int F^{c\,(a+b\,x)}\,\text{Tanh}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Tanh}[z]=\frac{-1+e^{2\,z}}{1+e^{2\,z}}$

Rule: If $n\in\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Tanh}[d+e\,x]^n\,dx\;\longrightarrow\;\int F^{c\,(a+b\,x)}\,\frac{\left(-1+e^{2\,(d+e\,x)}\right)^n}{\left(1+e^{2\,(d+e\,x)}\right)^n}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Tanh[d_.+e_.*x_]^n_.,x_Symbol] :=
  Int[ExpandIntegrand[F^(c*(a+b*x))*(-1+E^(2*(d+e*x)))^n/(1+E^(2*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Coth[d_.+e_.*x_]^n_.,x_Symbol] :=
  Int[ExpandIntegrand[F^(c*(a+b*x))*(1+E^(2*(d+e*x)))^n/(-1+E^(2*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

3. $\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n\,dx$

1: $\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n\,dx$ when $e^2\,n^2-b^2\,c^2\,\mathrm{Log}[F]^2\neq 0\,\land\,n<-1$

Reference: CRC 552h inverted

Reference: CRC 551h inverted

Rule: If $e^2\,n^2-b^2\,c^2\,\mathrm{Log}[F]^2\neq 0\,\land\,n<-1$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n\,dx\,\longrightarrow$$
$$-\frac{b\,c\,\mathrm{Log}[F]\,F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n}{e^2\,n^2-b^2\,c^2\,\mathrm{Log}[F]^2}-\frac{e\,n\,F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^{n+1}\,\mathrm{Sinh}[d+e\,x]}{e^2\,n^2-b^2\,c^2\,\mathrm{Log}[F]^2}+\frac{e^2\,n\,(n+1)}{e^2\,n^2-b^2\,c^2\,\mathrm{Log}[F]^2}\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^{n+2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sech[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*(Sech[d+e*x]^n/(e^2*n^2-b^2*c^2*Log[F]^2)) -
  e*n*F^(c*(a+b*x))*Sech[d+e*x]^(n+1)*(Sinh[d+e*x]/(e^2*n^2-b^2*c^2*Log[F]^2)) +
  e^2*n*((n+1)/(e^2*n^2-b^2*c^2*Log[F]^2))*Int[F^(c*(a+b*x))*Sech[d+e*x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*n^2+b^2*c^2*Log[F]^2,0] && LtQ[n,-1]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csch[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*(Csch[d+e*x]^n/(e^2*n^2-b^2*c^2*Log[F]^2)) -
  e*n*F^(c*(a+b*x))*Csch[d+e*x]^(n+1)*(Cosh[d+e*x]/(e^2*n^2-b^2*c^2*Log[F]^2)) -
  e^2*n*((n+1)/(e^2*n^2-b^2*c^2*Log[F]^2))*Int[F^(c*(a+b*x))*Csch[d+e*x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*n^2+b^2*c^2*Log[F]^2,0] && LtQ[n,-1]
```

2: $\int F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^n\,dx$ when $e^2\,(n-2)^2-b^2\,c^2\,\text{Log}[F]^2==0\,\land\,n\neq 1\,\land\,n\neq 2$

Reference: CRC 552h with $e^2\,(n-2)^2-b^2\,c^2\,\text{Log}[F]^2==0$

Reference: CRC 551h with $e^2\,(n-2)^2-b^2\,c^2\,\text{Log}[F]^2==0$

Rule: If $e^2\,(n-2)^2-b^2\,c^2\,\text{Log}[F]^2==0\,\land\,n\neq 1\,\land\,n\neq 2$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^n\,dx\,\longrightarrow\,\frac{b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^{n-2}}{e^2\,(n-1)\,(n-2)}+\frac{F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^{n-1}\,\text{Sinh}[d+e\,x]}{e\,(n-1)}$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sech[d_.+e_.*x_]^n_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Sech[d+e*x]^(n-2)/(e^2*(n-1)*(n-2)) +
  F^(c*(a+b*x))*Sech[d+e*x]^(n-1)*Sinh[d+e*x]/(e*(n-1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[e^2*(n-2)^2-b^2*c^2*Log[F]^2,0] && NeQ[n,1] && NeQ[n,2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csch[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Csch[d+e*x]^(n-2)/(e^2*(n-1)*(n-2)) -
  F^(c*(a+b*x))*Csch[d+e*x]^(n-1)*Cosh[d+e*x]/(e*(n-1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[e^2*(n-2)^2-b^2*c^2*Log[F]^2,0] && NeQ[n,1] && NeQ[n,2]
```

3: $\int F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^n\,dx$ when $e^2\,(n-2)^2-b^2\,c^2\,\text{Log}[F]^2\neq 0\,\land\,n>1\,\land\,n\neq 2$

Reference: CRC 552h

Reference: CRC 551h

Rule: If $e^2\,(n-2)^2-b^2\,c^2\,\text{Log}[F]^2\neq 0\,\land\,n>1\,\land\,n\neq 2$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^n\,dx\,\longrightarrow$$

$$\frac{b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^{n-2}}{e^2\,(n-1)\,(n-2)}+\frac{F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^{n-1}\,\text{Sinh}[d+e\,x]}{e\,(n-1)}+\frac{e^2\,(n-2)^2-b^2\,c^2\,\text{Log}[F]^2}{e^2\,(n-1)\,(n-2)}\int F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^{n-2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sech[d_.+e_.*x_]^n_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Sech[d+e*x]^(n-2)/(e^2*(n-1)*(n-2)) +
  F^(c*(a+b*x))*Sech[d+e*x]^(n-1)*Sinh[d+e*x]/(e*(n-1)) +
  (e^2*(n-2)^2-b^2*c^2*Log[F]^2)/(e^2*(n-1)*(n-2))*Int[F^(c*(a+b*x))*Sech[d+e*x]^(n-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*(n-2)^2-b^2*c^2*Log[F]^2,0] && GtQ[n,1] && NeQ[n,2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csch[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Csch[d+e*x]^(n-2)/(e^2*(n-1)*(n-2)) -
  F^(c*(a+b*x))*Csch[d+e*x]^(n-1)*Cosh[d+e*x]/(e*(n-1)) -
  (e^2*(n-2)^2-b^2*c^2*Log[F]^2)/(e^2*(n-1)*(n-2))*Int[F^(c*(a+b*x))*Csch[d+e*x]^(n-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*(n-2)^2-b^2*c^2*Log[F]^2,0] && GtQ[n,1] && NeQ[n,2]
```

x: $\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\mathrm{Sech}[z]==\frac{2\,e^z}{1+e^{2\,z}}$

Basis: $\mathrm{Csch}[z]==\frac{2\,e^{-z}}{1-e^{-2\,z}}$

Rule: If $n\in\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n\,dx\;\longrightarrow\;2^n\int F^{c\,(a+b\,x)}\,\frac{e^{n\,(d+e\,x)}}{\left(1+e^{2\,(d+e\,x)}\right)^n}\,dx$$

Program code:

```
(* Int[F_^(c_.*(a_.+b_.*x_))*Sech[d_.+e_.*x_]^n_.,x_Symbol] :=
  2^n*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(n*(d+e*x))/(1+E^(2*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n] *)
```

```
(* Int[F_^(c_.*(a_.+b_.*x_))*Csch[d_.+e_.*x_]^n_.,x_Symbol] :=
  2^n*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(-n*(d+e*x))/(1-E^(-2*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n] *)
```

4: $\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}$

Rule: If $n\in\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sech}[d+e\,x]^n\,dx\;\longrightarrow\;\frac{2^n\,e^{n\,(d+e\,x)}\,F^{c\,(a+b\,x)}}{e\,n+b\,c\,\mathrm{Log}[F]}\,\mathrm{Hypergeometric2F1}\!\left[n,\;\frac{n}{2}+\frac{b\,c\,\mathrm{Log}[F]}{2\,e},\;1+\frac{n}{2}+\frac{b\,c\,\mathrm{Log}[F]}{2\,e},\;-e^{2\,(d+e\,x)}\right]$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sech[d_.+e_.*x_]^n_.,x_Symbol] :=
  2^n*E^(n*(d+e*x))*F^(c*(a+b*x))/(e*n+b*c*Log[F])*Hypergeometric2F1[n,n/2+b*c*Log[F]/(2*e),1+n/2+b*c*Log[F]/(2*e),-E^(2*(d+e*x))] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csch[d_.+e_.*x_]^n_.,x_Symbol] :=
  (-2)^n*E^(n*(d+e*x))*F^(c*(a+b*x))/(e*n+b*c*Log[F])*Hypergeometric2F1[n,n/2+b*c*Log[F]/(2*e),1+n/2+b*c*Log[F]/(2*e),E^(2*(d+e*x))] /
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

5: $\int F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(1+e^{2\,(d+e\,x)}\right)^n\,\text{Sech}[d+e\,x]^n}{e^{n\,(d+e\,x)}} == 0$

Rule: If $n\notin\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sech}[d+e\,x]^n\,dx \longrightarrow \frac{\left(1+e^{2\,(d+e\,x)}\right)^n\,\text{Sech}[d+e\,x]^n}{e^{n\,(d+e\,x)}}\int F^{c\,(a+b\,x)}\,\frac{e^{n\,(d+e\,x)}}{\left(1+e^{2\,(d+e\,x)}\right)^n}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sech[d_.+e_.*x_]^n_.,x_Symbol] :=
  (1+E^(2*(d+e*x)))^n*Sech[d+e*x]^n/E^(n*(d+e*x))*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(n*(d+e*x))/(1+E^(2*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && Not[IntegerQ[n]]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csch[d_.+e_.*x_]^n_.,x_Symbol] :=
  (1-E^(-2*(d+e*x)))^n*Csch[d+e*x]^n/E^(-n*(d+e*x))*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(-n*(d+e*x))/(1-E^(-2*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && Not[IntegerQ[n]]
```

4. $\int u\, F^{c\,(a+b\,x)}\, (f+g\,\text{Sinh}[d+e\,x])^n\, dx$ when $f^2+g^2=0$

1: $\int F^{c\,(a+b\,x)}\, (f+g\,\text{Sinh}[d+e\,x])^n\, dx$ when $f^2+g^2=0\ \wedge\ n\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $f^2+g^2=0$, then $f+g\,\text{Sinh}[z]=2\,f\,\text{Cosh}\left[\frac{z}{2}-\frac{f\,\pi}{4\,g}\right]^2$

Basis: If $f-g=0$, then $f+g\,\text{Cosh}[z]=2\,g\,\text{Cosh}\left[\frac{z}{2}\right]^2$

Basis: If $f+g=0$, then $f+g\,\text{Cosh}[z]=2\,g\,\text{Sinh}\left[\frac{z}{2}\right]^2$

Rule: If $f^2+g^2=0\ \wedge\ n\in\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\, (f+g\,\text{Sinh}[d+e\,x])^n\, dx \longrightarrow 2^n\, f^n \int F^{c\,(a+b\,x)}\, \text{Cosh}\left[\frac{d}{2}+\frac{e\,x}{2}-\frac{f\,\pi}{4\,g}\right]^{2\,n} dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*(f_+g_.*Sinh[d_.+e_.*x_])^n_.,x_Symbol] :=
  2^n*f^n*Int[F^(c*(a+b*x))*Cosh[d/2+e*x/2-f*Pi/(4*g)]^(2*n),x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f^2+g^2,0] && ILtQ[n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*(f_+g_.*Cosh[d_.+e_.*x_])^n_.,x_Symbol] :=
  2^n*g^n*Int[F^(c*(a+b*x))*Cosh[d/2+e*x/2]^(2*n),x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f-g,0] && ILtQ[n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*(f_+g_.*Cosh[d_.+e_.*x_])^n_.,x_Symbol] :=
  2^n*g^n*Int[F^(c*(a+b*x))*Sinh[d/2+e*x/2]^(2*n),x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f+g,0] && ILtQ[n,0]
```

2: $\int F^{c\,(a+b\,x)}\,\text{Cosh}[d+e\,x]^m\,(f+g\,\text{Sinh}[d+e\,x])^n\,dx$ when $f^2+g^2==0\,\land\,(m\mid n)\in\mathbb{Z}\,\land\,m+n==0$

Derivation: Algebraic simplification

Basis: If $f^2+g^2==0$, then $\frac{\text{Cosh}[z]}{f+g\,\text{Sinh}[z]}==\frac{1}{g}\,\text{Tanh}\left[\frac{z}{2}-\frac{f\,\pi}{4\,g}\right]$

Basis: If $f-g==0$, then $\frac{\text{Sinh}[z]}{f+g\,\text{Cosh}[z]}==\frac{1}{g}\,\text{Tanh}\left[\frac{z}{2}\right]$

Basis: If $f+g==0$, then $\frac{\text{Sinh}[z]}{f+g\,\text{Cosh}[z]}==\frac{1}{g}\,\text{Coth}\left[\frac{z}{2}\right]$

Rule: If $f^2+g^2==0\,\land\,(m\mid n)\in\mathbb{Z}\,\land\,m+n==0$, then

$$\int F^{c\,(a+b\,x)}\,\text{Cosh}[d+e\,x]^m\,(f+g\,\text{Sinh}[d+e\,x])^n\,dx\;\longrightarrow\;g^n\int F^{c\,(a+b\,x)}\,\text{Tanh}\left[\frac{d}{2}+\frac{e\,x}{2}-\frac{f\,\pi}{4\,g}\right]^m\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_]^m_.*(f_+g_.*Sinh[d_.+e_.*x_])^n_.,x_Symbol] :=
  g^n*Int[F^(c*(a+b*x))*Tanh[d/2+e*x/2-f*Pi/(4*g)]^m,x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f^2+g^2,0] && IntegersQ[m,n] && EqQ[m+n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^m_.*(f_+g_.*Cosh[d_.+e_.*x_])^n_.,x_Symbol] :=
  g^n*Int[F^(c*(a+b*x))*Tanh[d/2+e*x/2]^m,x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f-g,0] && IntegersQ[m,n] && EqQ[m+n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^m_.*(f_+g_.*Cosh[d_.+e_.*x_])^n_.,x_Symbol] :=
  g^n*Int[F^(c*(a+b*x))*Coth[d/2+e*x/2]^m,x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f+g,0] && IntegersQ[m,n] && EqQ[m+n,0]
```

3: $\int F^{c\,(a+b\,x)}\,\frac{h+i\,\mathrm{Cosh}[d+e\,x]}{f+g\,\mathrm{Sinh}[d+e\,x]}\,dx$ when $f^2+g^2==0\land h^2-i^2==0\land g\,h+f\,i==0$

Derivation: Algebraic simplification

Basis: $\frac{h+i\,\mathrm{Cos}[z]}{f+g\,\mathrm{Sin}[z]}==\frac{2\,i\,\mathrm{Cos}[z]}{f+g\,\mathrm{Sin}[z]}+\frac{h-i\,\mathrm{Cos}[z]}{f+g\,\mathrm{Sin}[z]}$

Rule: If $f^2+g^2==0\land h^2-i^2==0\land g\,h+f\,i==0$, then

$$\int F^{c\,(a+b\,x)}\,\frac{h+i\,\mathrm{Cosh}[d+e\,x]}{f+g\,\mathrm{Sinh}[d+e\,x]}\,dx\,\longrightarrow\,2\,i\int F^{c\,(a+b\,x)}\,\frac{\mathrm{Cosh}[d+e\,x]}{f+g\,\mathrm{Sinh}[d+e\,x]}\,dx+\int F^{c\,(a+b\,x)}\,\frac{h-i\,\mathrm{Cosh}[d+e\,x]}{f+g\,\mathrm{Sinh}[d+e\,x]}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*(h_+i_.*Cosh[d_.+e_.*x_])/(f_+g_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  2*i*Int[F^(c*(a+b*x))*(Cosh[d+e*x]/(f+g*Sinh[d+e*x])),x] +
  Int[F^(c*(a+b*x))*((h-i*Cosh[d+e*x])/(f+g*Sinh[d+e*x])),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,i},x] && EqQ[f^2+g^2,0] && EqQ[h^2-i^2,0] && EqQ[g*h-f*i,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*(h_+i_.*Sinh[d_.+e_.*x_])/(f_+g_.*Cosh[d_.+e_.*x_]),x_Symbol] :=
  2*i*Int[F^(c*(a+b*x))*(Sinh[d+e*x]/(f+g*Cosh[d+e*x])),x] +
  Int[F^(c*(a+b*x))*((h-i*Sinh[d+e*x])/(f+g*Cosh[d+e*x])),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,i},x] && EqQ[f^2-g^2,0] && EqQ[h^2+i^2,0] && EqQ[g*h+f*i,0]
```

5: $\int F^{c\,u}\,\text{Hyper}[v]^n\,dx$ when $u == a+b\,x \land v == d+e\,x$

Derivation: Algebraic normalization

Rule: If $u == a+b\,x \land v == d+e\,x$, then

$$\int F^{c\,u}\,\text{Hyper}[v]^n\,dx \longrightarrow \int F^{c\,(a+b\,x)}\,\text{Hyper}\left[d+e\,x\right]^n\,dx$$

Program code:

```
Int[F_^(c_.*u_)*G_[v_]^n_.,x_Symbol] :=
  Int[F^(c*ExpandToSum[u,x])*G[ExpandToSum[v,x]]^n,x] /;
FreeQ[{F,c,n},x] && HyperbolicQ[G] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

6. $\int (f\,x)^m F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}^+$

1: $\int (f\,x)^m F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}^+\land m>0$

Derivation: Integration by parts

Note: Each term of the resulting integrand will be similar in form to the original integrand, but the degree of the monomial will be smaller by one.

Rule: If $n\in\mathbb{Z}^+\land m>0$, let $u=\int F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx$, then

$$\int (f\,x)^m F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]^n\,dx\;\longrightarrow\;(f\,x)^m u-f\,m\int (f\,x)^{m-1}u\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^n_.,x_Symbol] :=
  Module[{u=IntHide[F^(c*(a+b*x))*Sinh[d+e*x]^n,x]},
  Dist[(f*x)^m,u,x] - f*m*Int[(f*x)^(m-1)*u,x]] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] && GtQ[m,0]
```

```
Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_]^n_.,x_Symbol] :=
  Module[{u=IntHide[F^(c*(a+b*x))*Cosh[d+e*x]^n,x]},
  Dist[(f*x)^m,u,x] - f*m*Int[(f*x)^(m-1)*u,x]] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] && GtQ[m,0]
```

2: $\int (f\,x)^m\,F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]\,dx$ when $m<-1$

Derivation: Integration by parts

Basis: $(f\,x)^m = \partial_x \frac{(f\,x)^{m+1}}{f\,(m+1)}$

Basis: $\partial_x \left(F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]\right) = e\,F^{c\,(a+b\,x)}\,\text{Cosh}[d+e\,x] + b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]$

Rule: If $m<-1$, then

$$\int (f\,x)^m\,F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]\,dx \to$$
$$\frac{(f\,x)^{m+1}}{f\,(m+1)}\,F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x] - \frac{e}{f\,(m+1)}\int (f\,x)^{m+1}\,F^{c\,(a+b\,x)}\,\text{Cosh}[d+e\,x]\,dx - \frac{b\,c\,\text{Log}[F]}{f\,(m+1)}\int (f\,x)^{m+1}\,F^{c\,(a+b\,x)}\,\text{Sinh}[d+e\,x]\,dx$$

Program code:

```
Int[(f_.*x_)^m_*F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_],x_Symbol] :=
  (f*x)^(m+1)/(f*(m+1))*F^(c*(a+b*x))*Sinh[d+e*x] -
  e/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Cosh[d+e*x],x] -
  b*c*Log[F]/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Sinh[d+e*x],x] /;
FreeQ[{F,a,b,c,d,e,f,m},x] && (LtQ[m,-1] || SumSimplerQ[m,1])
```

```
Int[(f_.*x_)^m_*F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_],x_Symbol] :=
  (f*x)^(m+1)/(f*(m+1))*F^(c*(a+b*x))*Cosh[d+e*x] -
  e/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Sinh[d+e*x],x] -
  b*c*Log[F]/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Cosh[d+e*x],x] /;
FreeQ[{F,a,b,c,d,e,f,m},x] && (LtQ[m,-1] || SumSimplerQ[m,1])
```

x: $\int (f\,x)^m F^{c\,(a+b\,x)}\,\mathrm{Sinh}[d+e\,x]^n\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\mathrm{Sinh}[z] == -\frac{1}{2}\left(e^{-z}-e^{z}\right)$

Basis: $\mathrm{Cosh}[z] == \frac{1}{2}\left(e^{-z}+e^{z}\right)$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int (f\,x)^m F^{c\,(a+b\,x)}\,\mathrm{Sinh}[d+e\,x]^n\,dx \longrightarrow \frac{(-1)^n}{2^n}\int (f\,x)^m F^{c\,(a+b\,x)}\,\mathrm{ExpandIntegrand}\left[\left(e^{-(d+e\,x)}-e^{d+e\,x}\right)^n,\,x\right]dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^n_.,x_Symbol] :=
  (-1)^n/2^n*Int[ExpandIntegrand[(f*x)^m*F^(c*(a+b*x)),(E^(-(d+e*x))-E^(d+e*x))^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] *)
```

```
(* Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Cosh[d_.+e_.*x_]^n_.,x_Symbol] :=
  1/2^n*Int[ExpandIntegrand[(f*x)^m*F^(c*(a+b*x)),(E^(-(d+e*x))+E^(d+e*x))^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] *)
```

7. $\int u\, F^{c\,(a+b\,x)}\, \text{Sinh}[d+e\,x]^m\, \text{Cosh}[f+g\,x]^n\, dx$

1: $\int F^{c\,(a+b\,x)}\, \text{Sinh}[d+e\,x]^m\, \text{Cosh}[f+g\,x]^n\, dx$ when $m \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+$, then

$$\int F^{c\,(a+b\,x)}\, \text{Sinh}[d+e\,x]^m\, \text{Cosh}[f+g\,x]^n\, dx \,\to\, \int F^{c\,(a+b\,x)}\, \text{TrigReduce}\left[\text{Sinh}[d+e\,x]^m\, \text{Cosh}[f+g\,x]^n\right] dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^m_.*Cosh[f_.+g_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[F^(c*(a+b*x)),Sinh[d+e*x]^m*Cosh[f+g*x]^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && IGtQ[m,0] && IGtQ[n,0]
```

2: $\int x^p\, F^{c\,(a+b\,x)}\, \text{Sinh}[d+e\,x]^m\, \text{Cosh}[f+g\,x]^n\, dx$ when $m \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+$, then

$$\int x^p\, F^{c\,(a+b\,x)}\, \text{Sinh}[d+e\,x]^m\, \text{Cosh}[f+g\,x]^n\, dx \,\to\, \int x^p\, F^{c\,(a+b\,x)}\, \text{TrigReduce}\left[\text{Sinh}[d+e\,x]^m\, \text{Cosh}[f+g\,x]^n\right] dx$$

Program code:

```
Int[x_^p_.*F_^(c_.*(a_.+b_.*x_))*Sinh[d_.+e_.*x_]^m_.*Cosh[f_.+g_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^p*F^(c*(a+b*x)),Sinh[d+e*x]^m*Cosh[f+g*x]^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

8: $\int F^{c\,(a+b\,x)}\,\text{Hyper}[d+e\,x]^m\,\text{Hyper}[d+e\,x]^n\,dx$ when $m\in\mathbb{Z}^+\land n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m\in\mathbb{Z}^+\land n\in\mathbb{Z}^+$, then

$$\int F^{c\,(a+b\,x)}\,\text{Hyper}[d+e\,x]^m\,\text{Hyper}[d+e\,x]^n\,dx\,\rightarrow\,\int F^{c\,(a+b\,x)}\,\text{TrigToExp}\left[\text{Hyper}[d+e\,x]^m\,\text{Hyper}[d+e\,x]^n,\,x\right]dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*G_[d_.+e_.*x_]^m_.*H_[d_.+e_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^(c*(a+b*x)),G[d+e*x]^m*H[d+e*x]^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IGtQ[m,0] && IGtQ[n,0] && HyperbolicQ[G] && HyperbolicQ[H]
```

9: $\int F^{a+b\,x+c\,x^2}\,\text{Sinh}[d+e\,x+f\,x^2]^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int F^{a+b\,x+c\,x^2}\,\text{Sinh}[d+e\,x+f\,x^2]^n\,dx\,\rightarrow\,\int F^{a+b\,x+c\,x^2}\,\text{TrigToExp}\left[\text{Sinh}[d+e\,x+f\,x^2]^n\right]dx$$

Program code:

```
Int[F_^u_*Sinh[v_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^u,Sinh[v]^n,x],x] /;
FreeQ[F,x] && (LinearQ[u,x] || PolyQ[u,x,2]) && (LinearQ[v,x] || PolyQ[v,x,2]) && IGtQ[n,0]
```

```
Int[F_^u_*Cosh[v_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^u,Cosh[v]^n,x],x] /;
FreeQ[F,x] && (LinearQ[u,x] || PolyQ[u,x,2]) && (LinearQ[v,x] || PolyQ[v,x,2]) && IGtQ[n,0]
```

10: $\int F^{a+b\,x+c\,x^2}\,\text{Sinh}\left[d+e\,x+f\,x^2\right]^m\,\text{Cosh}\left[d+e\,x+f\,x^2\right]^n\,dx$ when $(m\mid n)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}^+$, then

$$\int F^{a+b\,x+c\,x^2}\,\text{Sinh}\left[d+e\,x+f\,x^2\right]^m\,\text{Cosh}\left[d+e\,x+f\,x^2\right]^n\,dx\;\longrightarrow\;\int F^{a+b\,x+c\,x^2}\,\text{TrigToExp}\left[\text{Sinh}\left[d+e\,x+f\,x^2\right]^m\,\text{Cosh}\left[d+e\,x+f\,x^2\right]^n\right]\,dx$$

Program code:

```
Int[F_^u_*Sinh[v_]^m_.*Cosh[v_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^u,Sinh[v]^m*Cosh[v]^n,x],x] /;
FreeQ[F,x] && (LinearQ[u,x] || PolyQ[u,x,2]) && (LinearQ[v,x] || PolyQ[v,x,2]) && IGtQ[m,0] && IGtQ[n,0]
```