

Rules for miscellaneous integrands

Piecewise constant extraction integration rules

x: $\int u (c x^n)^p dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c x^n)^p}{x^{n p}} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int u (c x^n)^p dx \rightarrow \frac{c^{\text{FracPart}[p]} (c x^n)^{\text{FracPart}[p]}}{x^{n \text{FracPart}[p]}} \int u x^{n p} dx$$

Program code:

```
(* Int[u.*(c.*x_^n_)^p_,x_Symbol] :=  
  c^FracPart[p]*(c*x^n)^FracPart[p]/x^(n*FracPart[p])*Int[u*x^(n*p),x] /;  
FreeQ[{c,n,p},x] && Not[IntegerQ[p]] *)
```

1: $\int u (c (a + b x)^n)^p dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(c (a + b x)^n)^p}{(a + b x)^{n p}} == 0$$

$$\text{Basis: } \frac{(c (a + b x)^n)^p}{(a + b x)^{n p}} == \frac{c^{\text{IntPart}[p]} (c (a + b x)^n)^{\text{FracPart}[p]}}{(a + b x)^{n \text{FracPart}[p]}}$$

Rule: If $p \notin \mathbb{Z}$, then

$$\int u (c (a + b x)^n)^p dx \rightarrow \frac{c^{\text{IntPart}[p]} (c (a + b x)^n)^{\text{FracPart}[p]}}{(a + b x)^{n \text{FracPart}[p]}} \int u (a + b x)^{n p} dx$$

Program code:

```
Int[u*(c.*(a_.+b_.*x_)^n_)^p_,x_Symbol] :=
  c^IntPart[p]*(c*(a+b*x)^n)^FracPart[p]/(a+b*x)^(n*FracPart[p])*Int[u*(a+b*x)^(n*p),x] /;
FreeQ[{a,b,c,n,p},x] && Not[IntegerQ[p]] && Not[MatchQ[u, x^n1_.*v_. /; EqQ[n,n1+1]]]
```

2: $\int u \left(c \left(d \left(a + b x \right)^n \right)^p \right)^q dx$ when $p \notin \mathbb{Z} \wedge q \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c (d (a + b x)^n)^p)^q}{(a + b x)^{n p q}} == 0$

Note: This should be generalized for arbitrarily deep nesting of powers.

Rule: If $p \notin \mathbb{Z} \wedge q \notin \mathbb{Z}$, then

$$\int u \left(c \left(d \left(a + b x \right)^n \right)^p \right)^q dx \rightarrow \frac{(c (d (a + b x)^n)^p)^q}{(a + b x)^{n p q}} \int u (a + b x)^{n p q} dx$$

Program code:

```
Int[u_.*(c_.*(d_.*(a_+b_.*x_)) ^p_) ^q_,x_Symbol] :=
  (c*(d*(a+b*x)) ^p) ^q/(a+b*x)^(p*q)*Int[u*(a+b*x)^(p*q),x] /;
FreeQ[{a,b,c,d,p,q},x] && Not[IntegerQ[p]] && Not[IntegerQ[q]]
```

```
Int[u_.*(c_.*(d_.*(a_+b_.*x_) ^n_) ^p_) ^q_,x_Symbol] :=
  (c*(d*(a+b*x) ^n) ^p) ^q/(a+b*x)^(n*p*q)*Int[u*(a+b*x)^(n*p*q),x] /;
FreeQ[{a,b,c,d,n,p,q},x] && Not[IntegerQ[p]] && Not[IntegerQ[q]]
```

Substitution integration rules

1: $\int \frac{\left(a + b F \left[c \frac{\sqrt{d+ex}}{\sqrt{f+gx}} \right] \right)^n}{A + Bx + Cx^2} dx$ when $Cdf - Aeg == 0 \wedge Beg - C(e f + dg) == 0 \wedge n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] == 2 (e f - d g) \text{Subst} \left[\frac{x}{(e - g x^2)^2} F \left[-\frac{d - f x^2}{e - g x^2} \right], x, \frac{\sqrt{d+ex}}{\sqrt{f+gx}} \right] \partial_x \frac{\sqrt{d+ex}}{\sqrt{f+gx}}$

Basis: If $C d f - A e g == 0 \wedge B e g - C (e f + d g) == 0$, then

$$\frac{1}{A+Bx+Cx^2} == \frac{2eg}{C(e f - d g)} \text{Subst} \left[\frac{1}{x}, x, \frac{\sqrt{d+ex}}{\sqrt{f+gx}} \right] \partial_x \frac{\sqrt{d+ex}}{\sqrt{f+gx}}$$

Rule: If $C d f - A e g == 0 \wedge B e g - C (e f + d g) == 0 \wedge n \in \mathbb{Z}^+$, then

$$\int \frac{\left(\frac{a + b F \left[c \frac{\sqrt{d+ex}}{\sqrt{f+gx}} \right]}{A + Bx + Cx^2} \right)^n dx \rightarrow \frac{2eg}{C(e f - d g)} \text{Subst} \left[\int \frac{(a + b F[cx])^n}{x} dx, x, \frac{\sqrt{d+ex}}{\sqrt{f+gx}} \right]$$

Program code:

```
Int[(a_+b_*F[c_*Sqrt[d_+e_*x_]/Sqrt[f_+g_*x_]])^n_/(A_+B_*x_+C_*x_^2),x_Symbol] :=
  2*e*g/(C*(e*f-d*g))*Subst[Int[(a+b*F[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,B,C,F},x] && EqQ[C*d*f-A*e*g,0] && EqQ[B*e*g-C*(e*f+d*g),0] && IGtQ[n,0]
```

```
Int[(a_+b_*F[c_*Sqrt[d_+e_*x_]/Sqrt[f_+g_*x_]])^n_/(A_+C_*x_^2),x_Symbol] :=
  2*e*g/(C*(e*f-d*g))*Subst[Int[(a+b*F[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,C,F},x] && EqQ[C*d*f-A*e*g,0] && EqQ[e*f+d*g,0] && IGtQ[n,0]
```

2: $\int \frac{\left(a + b F\left[c \frac{\sqrt{d+ex}}{\sqrt{f+gx}}\right]\right)^n}{A + Bx + Cx^2} dx$ when $Cdf - Aeg = 0 \wedge Beg - C(e f + dg) = 0 \wedge n \notin \mathbb{Z}^+$

Rule: If $Cdf - Aeg = 0 \wedge Beg - C(e f + dg) = 0 \wedge n \notin \mathbb{Z}^+$, then

$$\int \frac{\left(a + b F\left[c \frac{\sqrt{d+ex}}{\sqrt{f+gx}}\right]\right)^n}{A + Bx + Cx^2} dx \rightarrow \int \frac{\left(a + b F\left[c \frac{\sqrt{d+ex}}{\sqrt{f+gx}}\right]\right)^n}{A + Bx + Cx^2} dx$$

Program code:

```
Int[(a_.+b_.*F[c_.*Sqrt[d_.+e_.**x_]/Sqrt[f_.+g_.**x_]])^n/(A_.+B_.**x_+C_.**x_^2),x_Symbol] :=
  Unintegrable[(a+b*F[c*Sqrt[d+e*x]/Sqrt[f+g*x]])^n/(A+B*x+C*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,C,F,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[B*e*g-C*(e*f+d*g),0] && Not[IGtQ[n,0]]
```

```
Int[(a_.+b_.*F[c_.*Sqrt[d_.+e_.**x_]/Sqrt[f_.+g_.**x_]])^n/(A_.+C_.**x_^2),x_Symbol] :=
  Unintegrable[(a+b*F[c*Sqrt[d+e*x]/Sqrt[f+g*x]])^n/(A+C*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g,A,C,F,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[e*f+d*g,0] && Not[IGtQ[n,0]]
```

Derivative divides integration rules

$$1: \int \frac{y'[x]}{y[x]} dx$$

Reference: G&R 2.111.1.2, CRC 27, A&S 3.3.15

Derivation: Integration by substitution and reciprocal rule for integration

Note: Although powerful, this rule is not tried earlier because it is inefficient.

Rule:

$$\int \frac{y'[x]}{y[x]} dx \rightarrow \text{Log}[y[x]]$$

Program code:

```
Int[u_/y_,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Log[RemoveContent[y,x]] /;
    Not[FalseQ[q]]]
```

```
Int[u_/(y_*w_),x_Symbol] :=
  With[{q=DerivativeDivides[y*w,u,x]},
    q*Log[RemoveContent[y*w,x]] /;
    Not[FalseQ[q]]]
```

2: $\int y'[x] y[x]^m dx$ when $m \neq -1$

Reference: G&R 2.111.1.1, CRC 23, A&S 3.3.14

Derivation: Integration by substitution and power rule for integration

Note: Although powerful, this rule is not tried earlier because it is inefficient.

Rule: If $m \neq -1$, then

$$\int y'[x] y[x]^m dx \rightarrow \frac{y[x]^{m+1}}{m+1}$$

Program code:

```
Int[u_*y_^m_.,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*y^(m+1)/(m+1) /;
    Not[FalseQ[q]] /;
    FreeQ[m,x] && NeQ[m,-1]
```

```
Int[u_*y_^m_.*z_^n_.,x_Symbol] :=
  With[{q=DerivativeDivides[y*z,u*z^(n-m),x]},
    q*y^(m+1)*z^(n-m)/(m+1) /;
    Not[FalseQ[q]] /;
    FreeQ[{m,n},x] && NeQ[m,-1]
```

Algebraic simplification integration rules

1: $\int u \, dx$ when `SimplerIntegrandQ[SimplifyIntegrand[u, x], u, x]`

Derivation: Algebraic simplification

Rule: Let $v = \text{SimplifyIntegrand}[u, x]$, if `SimplerIntegrandQ[v, u, x]`, then

$$\int u \, dx \rightarrow \int v \, dx$$

Program code:

```
Int[u_, x_Symbol] :=
  With[{v=SimplifyIntegrand[u,x]},
    Int[v,x] /;
    SimplerIntegrandQ[v,u,x]
```


$$2. \int u \left(e \sqrt{a + b x^n} + f \sqrt{c + d x^n} \right)^m dx \text{ when } m \in \mathbb{Z}^-$$

$$1: \int u \left(e \sqrt{a + b x^n} + f \sqrt{c + d x^n} \right)^m dx \text{ when } m \in \mathbb{Z}^- \wedge b e^2 = d f^2$$

Derivation: Algebraic simplification

$$\text{Basis: If } b e^2 = d f^2, \text{ then } \frac{1}{e \sqrt{a + b x^n} + f \sqrt{c + d x^n}} = \frac{e \sqrt{a + b x^n} - f \sqrt{c + d x^n}}{a e^2 - c f^2}$$

Rule: If $m \in \mathbb{Z}^- \wedge b e^2 = d f^2$, then

$$\int u \left(e \sqrt{a + b x^n} + f \sqrt{c + d x^n} \right)^m dx \rightarrow (a e^2 - c f^2)^m \int u \left(e \sqrt{a + b x^n} - f \sqrt{c + d x^n} \right)^{-m} dx$$

Program code:

```
Int[u_.*(e_.*Sqrt[a_+b_.*x_^n_.]+f_.*Sqrt[c_+d_.*x_^n_.])^m_,x_Symbol] :=
  (a*e^2-c*f^2)^m*Int[ExpandIntegrand[u*(e*Sqrt[a+b*x^n]-f*Sqrt[c+d*x^n])^(-m),x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && ILtQ[m,0] && EqQ[b*e^2-d*f^2,0]
```

$$\mathbf{2:} \int u \left(e \sqrt{a + b x^n} + f \sqrt{c + d x^n} \right)^m dx \text{ when } m \in \mathbb{Z}^- \wedge a e^2 = c f^2$$

Derivation: Algebraic simplification

$$\text{Basis: If } a e^2 = c f^2, \text{ then } \frac{1}{e \sqrt{a + b z} + f \sqrt{c + d z}} = \frac{e \sqrt{a + b z} - f \sqrt{c + d z}}{(b e^2 - d f^2) z}$$

Rule: If $m \in \mathbb{Z}^- \wedge a e^2 = c f^2$, then

$$\int u \left(e \sqrt{a + b x^n} + f \sqrt{c + d x^n} \right)^m dx \rightarrow (b e^2 - d f^2)^m \int u x^{m n} \left(e \sqrt{a + b x^n} - f \sqrt{c + d x^n} \right)^{-m} dx$$

Program code:

```
Int[u_.*(e_.*Sqrt[a_+b_.*x_^n_.]+f_.*Sqrt[c_+d_.*x_^n_.])^m_,x_Symbol] :=
  (b*e^2-d*f^2)^m*Int[ExpandIntegrand[u*x^(m*n)*(e*Sqrt[a+b*x^n]-f*Sqrt[c+d*x^n])^(-m),x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && ILtQ[m,0] && EqQ[a*e^2-c*f^2,0]
```

3: $\int u^m (a u^n + v)^p w dx$ when $p \in \mathbb{Z} \wedge n \neq 0$

Derivation: Algebraic simplification

Basis: If $p \in \mathbb{Z}$, then $(a u^n + v)^p = u^{np} (a + u^{-n} v)^p$

Rule: If $p \in \mathbb{Z} \wedge n \neq 0$, then

$$\int u^m (a u^n + v)^p w dx \rightarrow \int u^{m+np} (a + u^{-n} v)^p w dx$$

Program code:

```
Int[u_^m_.*(a_.*u_^n_+v_)^p_.*w_,x_Symbol] :=
  Int[u^(m+n*p)*(a+u^(-n)*v)^p*w,x] /;
  FreeQ[{a,m,n},x] && IntegerQ[p] && Not[GtQ[n,0]] && Not[FreeQ[v,x]]
```

Derivative divides integration rules

$$\mathbf{1:} \int y'[x] (a + b y[x])^m (c + d y[x])^n dx$$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (a + b y[x])^m (c + d y[x])^n dx \rightarrow \text{Subst} \left[\int (a + b x)^m (c + d x)^n dx, x, y[x] \right]$$

Program code:

```
Int[u_*(a_.+b_.*y_)^m_.*(c_.+d_.*v_)^n_.,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Subst[Int[(a+b*x)^m*(c+d*x)^n,x],x,y] /;
    Not[FalseQ[q]] /;
    FreeQ[{a,b,c,d,m,n},x] && EqQ[v,y]
```

2: $\int y'[x] (a + b y[x])^m (c + d y[x])^n (e + f y[x])^p dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (a + b y[x])^m (c + d y[x])^n (e + f y[x])^p dx \rightarrow \text{Subst}\left[\int (a + b x)^m (c + d x)^n (e + f x)^p dx, x, y[x]\right]$$

Program code:

```
Int[u_*(a_.*b_.*y_)^m_.*(c_.*d_.*v_)^n_.*(e_.*f_.*w_)^p_.,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Subst[Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p,x],x,y] /;
    Not[FalseQ[q]] /;
    FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[v,y] && EqQ[w,y]
```

3: $\int y'[x] (a + b y[x])^m (c + d y[x])^n (e + f y[x])^p (g + h y[x])^q dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (a + b y[x])^m (c + d y[x])^n (e + f y[x])^p (g + h y[x])^q dx \rightarrow \text{Subst}\left[\int (a + b x)^m (c + d x)^n (e + f x)^p (g + h x)^q dx, x, y[x]\right]$$

Program code:

```
Int[u_*(a_.*b_.*y_)^m_.*(c_.*d_.*v_)^n_.*(e_.*f_.*w_)^p_.*(g_.*h_.*z_)^q_.,x_Symbol] :=
  With[{r=DerivativeDivides[y,u,x]},
    r*Subst[Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x],x,y] /;
    Not[FalseQ[r]] /;
    FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x] && EqQ[v,y] && EqQ[w,y] && EqQ[z,y]
```

4: $\int y'[x] (a + b y[x]^n)^p dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (a + b y[x]^n)^p dx \rightarrow \text{Subst}\left[\int (a + b x^n)^p dx, x, y[x]\right]$$

Program code:

```
Int[u_.*(a_+b_.*y_^n_),x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    a*Int[u,x] + b*q*Subst[Int[x^n,x],x,y] /;
    Not[FalseQ[q]]] /;
  FreeQ[{a,b,n},x]
```

```
Int[u_.*(a_+b_.*y_^n_)^p_,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Subst[Int[(a+b*x^n)^p,x],x,y] /;
    Not[FalseQ[q]]] /;
  FreeQ[{a,b,n,p},x]
```

5: $\int y'[x] y[x]^m (a + b y[x]^n)^p dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] y[x]^m (a + b y[x]^n)^p dx \rightarrow \text{Subst}\left[\int x^m (a + b x^n)^p dx, x, y[x]\right]$$

Program code:

```
Int[u_.*v_^m_.*(a_.*b_.*y_^n_)^p_.,x_Symbol] :=
  Module[{q,r},
    q*r*Subst[Int[x^m*(a+b*x^n)^p,x],x,y] /;
    Not[FalseQ[r=Divides[y^m,v^m,x]]] && Not[FalseQ[q=DerivativeDivides[y,u,x]]] /;
    FreeQ[{a,b,m,n,p},x]
```


6: $\int y'[x] (a + b y[x]^n + c y[x]^{2n})^p dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (a + b y[x]^n + c y[x]^{2n})^p dx \rightarrow \text{Subst}\left[\int (a + b x^n + c x^{2n})^p dx, x, y[x]\right]$$

Program code:

```
Int[u_.*(a_.+b_.*y_^n_+c_.*v_^n2_.)^p_,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Subst[Int[(a+b*x^n+c*x^(2*n))^p,x],x,y] /;
    Not[FalseQ[q]] /;
    FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && EqQ[v,y]
```

7: $\int y'[x] (A + B y[x]^n) (a + b y[x]^n + c y[x]^{2n})^p dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (A + B y[x]^n) (a + b y[x]^n + c y[x]^{2n})^p dx \rightarrow \text{Subst}\left[\int (A + B x^n) (a + b x^n + c x^{2n})^p dx, x, y[x]\right]$$

Program code:

```
Int[u_.*(A_+B_.*y_^n_)(a_+b_.*v_^n_+c_.*w_^n2_.)^p_,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Subst[Int[(A+B*x^n)*(a+b*x^n+c*x^(2*n))^p,x],x,y] /;
    Not[FalseQ[q]]] /;
  FreeQ[{a,b,c,A,B,n,p},x] && EqQ[n2,2*n] && EqQ[v,y] && EqQ[w,y]
```

```
Int[u_.*(A_+B_.*y_^n_)(a_+c_.*w_^n2_.)^p_,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Subst[Int[(A+B*x^n)*(a+c*x^(2*n))^p,x],x,y] /;
    Not[FalseQ[q]]] /;
  FreeQ[{a,c,A,B,n,p},x] && EqQ[n2,2*n] && EqQ[w,y]
```

$$\mathbf{8:} \int y'[x] y[x]^m (a + b y[x]^n + c y[x]^{2n})^p dx$$

Derivation: Integration by substitution

Rule:

$$\int y'[x] y[x]^m (a + b y[x]^n + c y[x]^{2n})^p dx \rightarrow \text{Subst} \left[\int x^m (a + b x^n + c x^{2n})^p dx, x, y[x] \right]$$

Program code:

```
Int[u_.*v_^m_.*(a_.+b_.*y_^n_.+c_.*w_^n2_.)^p_.,x_Symbol] :=
  Module[{q,r},
    q*r*Subst[Int[x^m*(a+b*x^n+c*x^(2*n))^p,x],x,y] /;
    Not[FalseQ[r=Divides[y^m,v^m,x]]] && Not[FalseQ[q=DerivativeDivides[y,u,x]]] /;
    FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && EqQ[w,y]
```

9: $\int y'[x] y[x]^m (A + B y[x]^n) (a + b y[x]^n + c y[x]^{2n})^p dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] y[x]^m (A + B y[x]^n) (a + b y[x]^n + c y[x]^{2n})^p dx \rightarrow \text{Subst}\left[\int x^m (A + B x^n) (a + b x^n + c x^{2n})^p dx, x, y[x]\right]$$

Program code:

```
Int[u_.*z_^m_.*(A_+B_.*y_^n_)*(a_.+b_.*v_^n_+c_.*w_^n2_.)^p_,x_Symbol] :=
Module[{q,r},
  q*r*Subst[Int[x^m*(A+B*x^n)*(a+b*x^n+c*x^(2*n))^p,x],x,y] /;
  Not[FalseQ[r=Divides[y^m,z^m,x]]] && Not[FalseQ[q=DerivativeDivides[y,u,x]]] /;
  FreeQ[{a,b,c,A,B,m,n,p},x] && EqQ[n2,2*n] && EqQ[v,y] && EqQ[w,y]
```

```
Int[u_.*z_^m_.*(A_+B_.*y_^n_)*(a_.+c_.*w_^n2_.)^p_,x_Symbol] :=
Module[{q,r},
  q*r*Subst[Int[x^m*(A+B*x^n)*(a+c*x^(2*n))^p,x],x,y] /;
  Not[FalseQ[r=Divides[y^m,z^m,x]]] && Not[FalseQ[q=DerivativeDivides[y,u,x]]] /;
  FreeQ[{a,c,A,B,m,n,p},x] && EqQ[n2,2*n] && EqQ[w,y]
```

10: $\int y'[x] (a + b y[x]^n)^m (c + d y[x]^n)^p dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (a + b y[x]^n)^m (c + d y[x]^n)^p dx \rightarrow \text{Subst}\left[\int (a + b x^n)^m (c + d x^n)^p dx, x, y[x]\right]$$

Program code:

```
Int[u_.*(a_.*b_.*y_^n_)^m_.*(c_.*d_.*v_^n_)^p_.,x_Symbol] :=
  With[{q=DerivativeDivides[y,u,x]},
    q*Subst[Int[(a+b*x^n)^m*(c+d*x^n)^p,x],x,y] /;
    Not[FalseQ[q]] /;
    FreeQ[{a,b,c,d,m,n,p},x] && EqQ[v,y]
```

11: $\int y'[x] (a + b y[x]^n)^m (c + d y[x]^n)^p (e + f y[x]^n)^q dx$

Derivation: Integration by substitution

Rule:

$$\int y'[x] (a + b y[x]^n)^m (c + d y[x]^n)^p (e + f y[x]^n)^q dx \rightarrow \text{Subst}\left[\int (a + b x^n)^m (c + d x^n)^p (e + f x^n)^q dx, x, y[x]\right]$$

Program code:

```
Int[u_.*(a_.*b_.*y_^n_)^m_.*(c_.*d_.*v_^n_)^p_.*(e_.*f_.*w_^n_)^q_,x_Symbol] :=
  With[{r=DerivativeDivides[y,u,x]},
    r*Subst[Int[(a+b*x^n)^m*(c+d*x^n)^p*(e+f*x^n)^q,x],x,y] /;
    Not[FalseQ[r]] /;
    FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && EqQ[v,y] && EqQ[w,y]
```

12. $\int u F^v dx$

1: $\int u F^v dx$ when $\partial_x v = u$

Derivation: Integration by substitution

Rule: If $\partial_x v = u$, then

$$\int u F^v dx \rightarrow \frac{F^v}{\text{Log}[F]}$$

Program code:

```
Int[u_*F^v_,x_Symbol] :=
  With[{q=DerivativeDivides[v,u,x]},
    q*F^v/Log[F] /;
    Not[FalseQ[q]] /;
    FreeQ[F,x]
```

2: $\int u v^m F^v dx$ when $\partial_x v = u$

Derivation: Integration by substitution

Rule: If $\partial_x v = u$, then

$$\int u v^m F^v dx \rightarrow \text{Subst}\left[\int x^m F^x dx, x, v\right]$$

Program code:

```
Int[u_*w_^m_.*F^v_,x_Symbol] :=
  With[{q=DerivativeDivides[v,u,x]},
    q*Subst[Int[x^m*F^x,x],x,v] /;
    Not[FalseQ[q]] /;
    FreeQ[{F,m},x] && EqQ[w,v]
```


$$13. \int F[f[x]^p, g[x]^q] f[x]^r g[x]^s (c f'[x] g[x] + d f[x] g'[x]) dx$$

$$1: \int u (a + b v^p w^q)^m v^r w^s dx \text{ when } p(s+1) = q(r+1) \wedge r \neq -1 \wedge \frac{p}{r+1} \in \mathbb{Z} \wedge \text{FreeQ}\left[\frac{u}{p w \partial_x v + q v \partial_x w}, x\right]$$

Derivation: Integration by substitution

Basis: If $p(s+1) = q(r+1) \wedge r \neq -1 \wedge \frac{p}{r+1} \in \mathbb{Z}$, then

$$F[f[x]^p g[x]^q] f[x]^r g[x]^s (p g[x] f'[x] + q f[x] g'[x]) = \frac{p}{r+1} \text{Subst}\left[F\left[x^{\frac{p}{r+1}}\right], x, f[x]^{r+1} g[x]^{s+1}\right] \partial_x (f[x]^{r+1} g[x]^{s+1})$$

Rule: If $p(s+1) = q(r+1) \wedge r \neq -1 \wedge \frac{p}{r+1} \in \mathbb{Z}$, let $c = \frac{u}{p w \partial_x v + q v \partial_x w}$, if $\text{FreeQ}[c, x]$, then

$$\int u (a + b v^p w^q)^m v^r w^s dx \rightarrow \frac{c p}{r+1} \text{Subst}\left[\int (a + b x^{\frac{p}{r+1}})^m dx, x, v^{r+1} w^{s+1}\right]$$

Program code:

```
Int[u_*(a_+b_.*v_^p_.*w_^p_)^m_,x_Symbol] :=
  With[{c=Simplify[u/(w*D[v,x]+v*D[w,x])]},
    c*Subst[Int[(a+b*x^p)^m,x],x,v*w] /;
    FreeQ[c,x] /;
    FreeQ[{a,b,m,p},x] && IntegerQ[p]
```

```
Int[u_*(a_+b_.*v_^p_.*w_^q_)^m_.*v_^r_,x_Symbol] :=
  With[{c=Simplify[u/(p*w*D[v,x]+q*v*D[w,x])]},
    c*p/(r+1)*Subst[Int[(a+b*x^(p/(r+1)))^m,x],x,v^(r+1)*w] /;
    FreeQ[c,x] /;
    FreeQ[{a,b,m,p,q,r},x] && EqQ[p,q*(r+1)] && NeQ[r,-1] && IntegerQ[p/(r+1)]
```

```

Int[u_*(a_+b_.*v_^p_.*w_^q_.)^m_.*v_^r_.*w_^s_.,x_Symbol] :=
  With[{c=Simplify[u/(p*w*D[v,x]+q*v*D[w,x])]},
    c*p/(r+1)*Subst[Int[(a+b*x^(p/(r+1)))^m,x],x,v^(r+1)*w^(s+1)] /;
  FreeQ[c,x] /;
  FreeQ[{a,b,m,p,q,r,s},x] && EqQ[p*(s+1),q*(r+1)] && NeQ[r,-1] && IntegerQ[p/(r+1)]

```

2: $\int u (a v^p + b w^q)^m v^r w^s dx$ when $p(s+1) + q(m p + r + 1) = 0 \wedge s \neq -1 \wedge \frac{q}{s+1} \in \mathbb{Z} \wedge m \in \mathbb{Z} \wedge \text{FreeQ}\left[\frac{u}{p w \partial_x v - q v \partial_x w}, x\right]$

Derivation: Integration by substitution

Basis: If $p(s+1) + q(m p + r + 1) = 0 \wedge s+1 \neq 0 \wedge \frac{q}{s+1} \in \mathbb{Z} \wedge m \in \mathbb{Z}$, then

$$(a f[x]^p + b g[x]^q)^m f[x]^r g[x]^s (p g[x] f'[x] - q f[x] g'[x]) = -\frac{q}{s+1} \text{Subst}\left[\left(a + b x^{\frac{q}{s+1}}\right)^m, x, f[x]^{m p + r + 1} g[x]^{s+1}\right] \partial_x (f[x]^{m p + r + 1} g[x]^{s+1})$$

Rule: If $p(s+1) + q(m p + r + 1) = 0 \wedge s \neq -1 \wedge \frac{q}{s+1} \in \mathbb{Z} \wedge m \in \mathbb{Z}$, let $c = \frac{u}{p w \partial_x v - q v \partial_x w}$, if $\text{FreeQ}[c, x]$, then

$$\int u (a v^p + b w^q)^m v^r w^s dx \rightarrow -\frac{c q}{s+1} \text{Subst}\left[\int (a + b x^{\frac{q}{s+1}})^m dx, x, v^{m p + r + 1} w^{s+1}\right]$$

Program code:

```

Int[u_*(a_.*v_^p_.+b_.*w_^q_.)^m_.,x_Symbol] :=
  With[{c=Simplify[u/(p*w*D[v,x]-q*v*D[w,x])]},
    c*p*Subst[Int[(b+a*x^p)^m,x],x,v*w^(m*q+1)] /;
  FreeQ[c,x] /;
  FreeQ[{a,b,m,p,q},x] && EqQ[p+q*(m*p+1),0] && IntegerQ[p] && IntegerQ[m]

```

```

(* Int[u_*(a_.*v_^p_.+b_.*w_^q_.)^m_.,x_Symbol] :=
  With[{c=Simplify[u/(p*w*D[v,x]-q*v*D[w,x])]},
    -c*q*Subst[Int[(a+b*x^q)^m,x],x,v^(m*p+1)*w] /;
  FreeQ[c,x] /;
  FreeQ[{a,b,m,p,q},x] && EqQ[p+q*(m*p+1),0] && IntegerQ[q] && IntegerQ[m] *)

```

```

Int[u_*(a_.*v_^p_.+b_.*w_^q_.)^m_.*v_^r_. ,x_Symbol] :=
  With[{c=Simplify[u/(p*w*D[v,x]-q*v*D[w,x])]},
    -c*q*Subst[Int[(a+b*x^q)^m,x],x,v^(m*p+r+1)*w] /;
  FreeQ[c,x] /;
  FreeQ[{a,b,m,p,q,r},x] && EqQ[p+q*(m*p+r+1),0] && IntegerQ[q] && IntegerQ[m]

```

```

Int[u_*(a_.*v_^p_.+b_.*w_^q_.)^m_.*w_^s_. ,x_Symbol] :=
  With[{c=Simplify[u/(p*w*D[v,x]-q*v*D[w,x])]},
    -c*q/(s+1)*Subst[Int[(a+b*x^(q/(s+1)))^m,x],x,v^(m*p+1)*w^(s+1)] /;
  FreeQ[c,x] /;
  FreeQ[{a,b,m,p,q,s},x] && EqQ[p*(s+1)+q*(m*p+1),0] && NeQ[s,-1] && IntegerQ[q/(s+1)] && IntegerQ[m]

```

```

Int[u_*(a_.*v_^p_.+b_.*w_^q_.)^m_.*v_^r_.*w_^s_. ,x_Symbol] :=
  With[{c=Simplify[u/(p*w*D[v,x]-q*v*D[w,x])]},
    -c*q/(s+1)*Subst[Int[(a+b*x^(q/(s+1)))^m,x],x,v^(m*p+r+1)*w^(s+1)] /;
  FreeQ[c,x] /;
  FreeQ[{a,b,m,p,q,r,s},x] && EqQ[p*(s+1)+q*(m*p+r+1),0] && NeQ[s,-1] && IntegerQ[q/(s+1)] && IntegerQ[m]

```

Substitution integration rules

1: $\int x^m F[x^{m+1}] dx$ when $m \neq -1$

Derivation: Integration by substitution

Basis: If $m \neq -1$, then $x^m F[x^{m+1}] = \frac{1}{m+1} F[x^{m+1}] \partial_x x^{m+1}$

Rule: If $m \neq -1$, then

$$\int x^m F[x^{m+1}] dx \rightarrow \frac{1}{m+1} \text{Subst}\left[\int F[x] dx, x, x^{m+1}\right]$$

Program code:

```
Int[u_*x^m_,x_Symbol] :=
  1/(m+1)*Subst[Int[SubstFor[x^(m+1),u,x],x],x,x^(m+1)] /;
FreeQ[m,x] && NeQ[m,-1] && FunctionOfQ[x^(m+1),u,x]
```

2: $\int F[(a + b x)^{1/n}, x] dx$ when $n \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $F[(a + b x)^{1/n}, x] = \frac{n}{b} \text{Subst}\left[x^{n-1} F\left[x, -\frac{a}{b} + \frac{x^n}{b}\right], x, (a + b x)^{1/n}\right] \partial_x (a + b x)^{1/n}$

Rule: If $n \in \mathbb{Z}$, then

$$\int F[(a + b x)^{1/n}, x] dx \rightarrow \frac{n}{b} \text{Subst}\left[\int x^{n-1} F\left[x, -\frac{a}{b} + \frac{x^n}{b}\right] dx, x, (a + b x)^{1/n}\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_, x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfLinear[u,x]},
    ShowStep["", "Int[F[(a+b*x)^(1/n), x], x]",
      "n/b*Subst[Int[x^(n-1)*F[x, -a/b+x^n/b], x], x, (a+b*x)^(1/n)]", Hold[
        lst[[2]]*lst[[4]]*Subst[Int[lst[[1]], x], x, lst[[3]]^(1/lst[[2])]]] /;
    Not[FalseQ[lst]] && SubstForFractionalPowerQ[u, lst[[3]], x]] /;
SimplifyFlag,

Int[u_, x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfLinear[u,x]},
    lst[[2]]*lst[[4]]*Subst[Int[lst[[1]], x], x, lst[[3]]^(1/lst[[2])]] /;
    Not[FalseQ[lst]] && SubstForFractionalPowerQ[u, lst[[3]], x]]]
```

3: $\int F\left[\left(\frac{a + b x}{c + d x}\right)^{1/n}, x\right] dx$ when $n \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then

$$F\left[\left(\frac{a+bx}{c+dx}\right)^{1/n}, x\right] = n(bd - ac) \operatorname{Subst}\left[\frac{x^{n-1}}{(b-dx^n)^2} F\left[x, \frac{-a+cx^n}{b-dx^n}\right], x, \left(\frac{a+bx}{c+dx}\right)^{1/n}\right] \partial_x \left(\frac{a+bx}{c+dx}\right)^{1/n}$$

Rule: If $n \in \mathbb{Z}$, then

$$\int F\left[\left(\frac{a+bx}{c+dx}\right)^{1/n}, x\right] dx \rightarrow n(bd - ac) \operatorname{Subst}\left[\int \frac{x^{n-1}}{(b-dx^n)^2} F\left[x, \frac{-a+cx^n}{b-dx^n}\right] dx, x, \left(\frac{a+bx}{c+dx}\right)^{1/n}\right]$$

Program code:

```
If[TrueQ[LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfQuotientOfLinears[u,x]},
    ShowStep["", "Int[F[(a+b*x)/(c+d*x)^(1/n), x], x]",
      "n*(b*c-a*d)*Subst[Int[x^(n-1)*F[x, (-a+c*x^n)/(b-d*x^n)]/(b-d*x^n)^2, x], x, ((a+b*x)/(c+d*x)^(1/n))", Hold[
        lst[[2]]*lst[[4]]*Subst[Int[lst[[1]], x], x, lst[[3]]^(1/lst[[2]])]]] /;
    Not[FalseQ[lst]]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfQuotientOfLinears[u,x]},
    lst[[2]]*lst[[4]]*Subst[Int[lst[[1]], x], x, lst[[3]]^(1/lst[[2]])] /;
    Not[FalseQ[lst]]]
```

Piecewise constant extraction integration rules

$$\mathbf{1:} \int u (v^m w^n \dots)^p dx \text{ when } p \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(a F[x]^m G[x]^n \dots)^p}{F[x]^m G[x]^n \dots} == 0$$

$$\text{Basis: } \frac{(a v^m w^n \dots)^p}{v^m w^n \dots} == \frac{a^{\text{IntPart}[p]} (a v^m w^n \dots)^{\text{FracPart}[p]}}{v^{m \text{FracPart}[p]} w^{n \text{FracPart}[p]} \dots}$$

Rule: If $p \notin \mathbb{Z}$, then

$$\int u (a v^m w^n \dots)^p dx \rightarrow \frac{a^{\text{IntPart}[p]} (a v^m w^n \dots)^{\text{FracPart}[p]}}{v^{m \text{FracPart}[p]} w^{n \text{FracPart}[p]} \dots} \int u v^{m \text{FracPart}[p]} w^{n \text{FracPart}[p]} \dots dx$$

Program code:

```
Int[u_.*(a_.*v_^m_.*w_^n_.*z_^q_.)^p_,x_Symbol] :=
  a^IntPart[p]*(a*v^m*w^n*z^q)^FracPart[p]/(v^(m*FracPart[p])*w^(n*FracPart[p])*z^(q*FracPart[p]))*Int[u*v^(m*p)*w^(n*p)*z^(p*q),x] /;
FreeQ[{a,m,n,p,q},x] && Not[IntegerQ[p]] && Not[FreeQ[v,x]] && Not[FreeQ[w,x]] && Not[FreeQ[z,x]]
```

```
Int[u_.*(a_.*v_^m_.*w_^n_.)^p_,x_Symbol] :=
  a^IntPart[p]*(a*v^m*w^n)^FracPart[p]/(v^(m*FracPart[p])*w^(n*FracPart[p]))*Int[u*v^(m*p)*w^(n*p),x] /;
FreeQ[{a,m,n,p},x] && Not[IntegerQ[p]] && Not[FreeQ[v,x]] && Not[FreeQ[w,x]]
```

```
Int[u_.*(a_.*v_^m_.)^p_,x_Symbol] :=
  a^IntPart[p]*(a*v^m)^FracPart[p]/v^(m*FracPart[p])*Int[u*v^(m*p),x] /;
FreeQ[{a,m,p},x] && Not[IntegerQ[p]] && Not[FreeQ[v,x]] && Not[EqQ[a,1] && EqQ[m,1]] && Not[EqQ[v,x] && EqQ[m,1]]
```

2. $\int u (a + b v^n)^p dx$ when $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}^-$

1: $\int u (a + b x^n)^p dx$ when $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}^-$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a + b x^n)^p}{x^{np} \left(1 + \frac{a x^{-n}}{b}\right)^p} = 0$

Rule: If $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}^-$, then

$$\int u (a + b x^n)^p dx \rightarrow \frac{b^{\text{IntPart}[p]} (a + b x^n)^{\text{FracPart}[p]}}{x^{n \text{FracPart}[p]} \left(1 + \frac{a x^{-n}}{b}\right)^{\text{FracPart}[p]}} \int u x^{np} \left(1 + \frac{a x^{-n}}{b}\right)^p dx$$

Program code:

```
Int[u_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  b^IntPart[p]*(a+b*x^n)^FracPart[p]/(x^(n*FracPart[p])*(1+a*x^(-n)/b)^FracPart[p])*Int[u*x^(n*p)*(1+a*x^(-n)/b)^p,x] /;
FreeQ[{a,b,p},x] && Not[IntegerQ[p]] && ILtQ[n,0] && Not[RationalFunctionQ[u,x]] && IntegerQ[p+1/2]
```


2: $\int u (a + b v^n)^p dx$ when $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}^-$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+b F[x]^n)^p}{F[x]^{np} (b+a F[x]^{-n})^p} == 0$

Basis: $\frac{(a+b v^n)^p}{v^{np} (b+a v^{-n})^p} == \frac{(a+b v^n)^{\text{FracPart}[p]}}{v^{n \text{FracPart}[p]} (b+a v^{-n})^{\text{FracPart}[p]}}$

Rule: If $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}^-$, then

$$\int u (a + b v^n)^p dx \rightarrow \frac{(a + b v^n)^{\text{FracPart}[p]}}{v^{n \text{FracPart}[p]} (b + a v^{-n})^{\text{FracPart}[p]}} \int u v^{np} (b + a v^{-n})^p dx$$

Program code:

```
Int[u_.*(a_.*b_.*v_^n_)^p_,x_Symbol] :=
  (a+b*v^n)^FracPart[p]/(v^(n*FracPart[p])*(b+a*v^(-n))^FracPart[p])*Int[u*v^(n*p)*(b+a*v^(-n))^p,x] /;
FreeQ[{a,b,p},x] && Not[IntegerQ[p]] && ILtQ[n,0] && BinomialQ[v,x] && Not[LinearQ[v,x]]
```

3: $\int u (a + b x^m v^n)^p dx$ when $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}^-$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+b x^m F[x]^n)^p}{F[x]^{n p} (b x^m + a F[x]^{-n})^p} == 0$

Basis: $\frac{(a+b x^m v^n)^p}{v^{n p} (b x^m + a v^{-n})^p} == \frac{(a+b x^m v^n)^{\text{FracPart}[p]}}{v^{n \text{FracPart}[p]} (b x^m + a v^{-n})^{\text{FracPart}[p]}}$

Rule: If $p \notin \mathbb{Z} \wedge n \in \mathbb{Z}^-$, then

$$\int u (a + b x^m v^n)^p dx \rightarrow \frac{(a + b x^m v^n)^{\text{FracPart}[p]}}{v^{n \text{FracPart}[p]} (b x^m + a v^{-n})^{\text{FracPart}[p]}} \int u v^{n p} (b x^m + a v^{-n})^p dx$$

Program code:

```
Int[u_.*(a_.+b_.*x_^m_.*v_^n_)^p_,x_Symbol] :=
  (a+b*x^m*v^n)^FracPart[p]/(v^(n*FracPart[p])*(b*x^m+a*v^(-n))^FracPart[p])*Int[u*v^(n*p)*(b*x^m+a*v^(-n))^p,x] /;
FreeQ[{a,b,m,p},x] && Not[IntegerQ[p]] && ILtQ[n,0] && BinomialQ[v,x]
```

4: $\int u (a x^r + b x^s)^m dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a x^r + b x^s)^m}{x^{m r} (a + b x^{s-r})^m} == 0$

Basis: $\frac{(a x^r + b x^s)^m}{x^{r m} (a + b x^{s-r})^m} == \frac{(a x^r + b x^s)^{\text{FracPart}[m]}}{x^{r \text{FracPart}[m]} (a + b x^{s-r})^{\text{FracPart}[m]}}$

Note: This rule should be generalized to handle an arbitrary number of terms.

Rule: If $m \notin \mathbb{Z}$, then

$$\int u (a x^r + b x^s)^m dx \rightarrow \frac{(a x^r + b x^s)^{\text{FracPart}[m]}}{x^{r \text{FracPart}[m]} (a + b x^{s-r})^{\text{FracPart}[m]}} \int u x^{m r} (a + b x^{s-r})^m dx$$

Program code:

```
Int[u_*(a_.*x^r_.+b_.*x^s_.)^m_,x_Symbol] :=
  With[{v=(a*x^r+b*x^s)^FracPart[m]/(x^(r*FracPart[m])*(a+b*x^(s-r))^FracPart[m])},
    v*Int[u*x^(m*r)*(a+b*x^(s-r))^m,x] /;
    NeQ[Simplify[v],1] /;
    FreeQ[{a,b,m,r,s},x] && Not[IntegerQ[m]] && PosQ[s-r]
```

Algebraic expansion integration rules

1: $\int \frac{u}{a + b x^n} dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \frac{u}{a + b x^n} dx \rightarrow \int \text{RationalFunctionExpand}\left[\frac{u}{a + b x^n}, x\right] dx$$

Program code:

```
Int[u_/(a_+b_.*x^n_),x_Symbol] :=
  With[{v=RationalFunctionExpand[u/(a+b*x^n),x]},
    Int[v,x] /;
    SumQ[v] /;
    FreeQ[{a,b},x] && IGtQ[n,0]
```

$$2. \int u (a + b x^n + c x^{2n})^p dx$$

$$1. \int u (a + b x^n + c x^{2n})^p dx \text{ when } b^2 - 4ac = 0$$

$$1: \int u (a + b x^n + c x^{2n})^p dx \text{ when } b^2 - 4ac = 0 \wedge p \in \mathbb{Z}$$

Derivation: Algebraic simplification

Basis: If $b^2 - 4ac = 0$, then $a + bz + cz^2 = \frac{1}{4c} (b + 2cz)^2$

Rule: If $b^2 - 4ac = 0 \wedge p \in \mathbb{Z}$, then

$$\int u (a + b x^n + c x^{2n})^p dx \rightarrow \frac{1}{4^p c^p} \int u (b + 2cx^n)^{2p} dx$$

Program code:

```
Int[u*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/(4^p*c^p)*Int[u*(b+2*c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p] && Not[AlgebraicFunctionQ[u,x]]
```

2: $\int u (a + b x^n + c x^{2n})^p dx$ when $b^2 - 4ac = 0 \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

■ Basis: If $b^2 - 4ac = 0$, then $\partial_x \frac{(a + b x^n + c x^{2n})^p}{(b + 2c x^n)^{2p}} = 0$

Rule: If $b^2 - 4ac = 0 \wedge p \notin \mathbb{Z}$, then

$$\int u (a + b x^n + c x^{2n})^p dx \rightarrow \frac{(a + b x^n + c x^{2n})^p}{(b + 2c x^n)^{2p}} \int u (b + 2c x^n)^{2p} dx$$

Program code:

```
Int[u_*(a_.*b_.*x_^n_.*c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^p/(b+2*c*x^n)^(2*p)*Int[u*(b+2*c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && Not[AlgebraicFunctionQ[u,x]]
```

$$2. \int u (a + b x^n + c x^{2n})^p dx \text{ when } b^2 - 4ac \neq 0$$

$$1: \int \frac{u}{a + b x^n + c x^{2n}} dx \text{ when } n \in \mathbb{Z}^+$$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \frac{u}{a + b x^n + c x^{2n}} dx \rightarrow \int \text{RationalFunctionExpand}\left[\frac{u}{a + b x^n + c x^{2n}}, x\right] dx$$

Program code:

```
Int[u/(a_.+b_.*x_^n_.+c_.*x_^n2_.),x_Symbol] :=
  With[{v=RationalFunctionExpand[u/(a+b*x^n+c*x^(2*n)),x]},
    Int[v,x] /;
    SumQ[v] /;
    FreeQ[{a,b,c},x] && EqQ[n2,2*n] && IGtQ[n,0]
```

3: $\int \frac{u}{a x^m + b \sqrt{c x^n}} dx$

Derivation: Algebraic simplification

Basis: $\frac{1}{z+w} == \frac{z-w}{z^2-w^2}$

Rule:

$$\int \frac{u}{a x^m + b \sqrt{c x^n}} dx \rightarrow \int \frac{u (a x^m - b \sqrt{c x^n})}{a^2 x^{2m} - b^2 c x^n} dx$$

Program code:

```
Int[u_/(a_.*x_^m_.+b_.*Sqrt[c_.*x_^n_]),x_Symbol] :=
  Int[u*(a*x^m-b*Sqrt[c*x^n])/(a^2*x^(2*m)-b^2*c*x^n),x] /;
FreeQ[{a,b,c,m,n},x]
```

Substitution integration rules

$$1: \int F[a + b x] dx$$

Derivation: Integration by substitution

$$\text{Basis: } F[a + b x] = \frac{1}{b} F[a + b x] \partial_x (a + b x)$$

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfLinear[u,x]},
    ShowStep["",Int[F[a+b*x],x],"Subst[Int[F[x],x],x,a+b*x]/b",Hold[
      Dist[1/lst[[3]],Subst[Int[lst[[1]],x],x,lst[[2]]+lst[[3]]*x],x]] /;
    Not[FalseQ[lst]]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfLinear[u,x]},
    Dist[1/lst[[3]],Subst[Int[lst[[1]],x],x,lst[[2]]+lst[[3]]*x],x] /;
    Not[FalseQ[lst]]]
```

$$2. \int x^m F[x^n] dx \text{ when } \text{GCD}[m+1, n] > 1$$

$$1: \int \frac{F[(c x)^n]}{x} dx$$

Derivation: Integration by substitution

$$\text{Basis: } \frac{F[(c x)^n]}{x} = \frac{F[(c x)^n]}{n (c x)^n} \partial_x (c x)^n$$

Rule:

$$\int \frac{F[(c x)^n]}{x} dx \rightarrow \frac{1}{n} \text{Subst}\left[\int \frac{F[x]}{x} dx, x, (c x)^n\right]$$

Program code:

```

If[TrueQ[$LoadShowSteps],

Int[u_/x_,x_Symbol] :=
  With[{lst=PowerVariableExpn[u,0,x]},
    ShowStep["","Int[F[(c*x)^n]/x,x]","Subst[Int[F[x]/x,x],x,(c*x)^n]/n",Hold[
      1/lst[[2]]*Subst[Int[NormalizeIntegrand[Simplify[lst[[1]]/x],x],x],x,(lst[[3]]*x)^lst[[2]]]] /;
    Not[FalseQ[lst]] && NeQ[lst[[2]],0]] /;
  SimplifyFlag && NonsumQ[u] && Not[RationalFunctionQ[u,x]],

Int[u_/x_,x_Symbol] :=
  With[{lst=PowerVariableExpn[u,0,x]},
    1/lst[[2]]*Subst[Int[NormalizeIntegrand[Simplify[lst[[1]]/x],x],x],x,(lst[[3]]*x)^lst[[2]] /;
    Not[FalseQ[lst]] && NeQ[lst[[2]],0]] /;
  NonsumQ[u] && Not[RationalFunctionQ[u,x]]]

```

2: $\int x^m F[x^n] dx$ when $m \neq -1 \wedge \text{GCD}[m+1, n] > 1$

Derivation: Integration by substitution

Basis: Let $g = \text{GCD}[m+1, n]$, then $x^m F[x^n] = \frac{1}{g} (x^g)^{(m+1)/g-1} F[(x^g)^{n/g}] \partial_x x^g$

Rule: If $m \neq -1$, let $g = \text{GCD}[m+1, n]$, if $g > 1$, then

$$\int x^m F[x^n] dx \rightarrow \frac{1}{g} \text{Subst}\left[\int x^{(m+1)/g-1} F[x^{n/g}] dx, x, x^g\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_*x_^m_, x_Symbol] :=
  With[{lst=PowerVariableExpn[u,m+1,x]},
    ShowStep["If g=GCD[m+1,n]>1,", "Int[x^m*F[x^n],x]",
      "1/g*Subst[Int[x^((m+1)/g-1)*F[x^(n/g)],x],x,x^g]", Hold[
        1/lst[[2]]*Subst[Int[NormalizeIntegrand[Simplify[lst[[1]]/x],x],x],x,(lst[[3]]*x)^lst[[2]]] /;
        Not[FalseQ[lst]] && NeQ[lst[[2]],m+1]] /;
    SimplifyFlag && IntegerQ[m] && NeQ[m,-1] && NonsumQ[u] && (GtQ[m,0] || Not[AlgebraicFunctionQ[u,x]]),

Int[u_*x_^m_, x_Symbol] :=
  With[{lst=PowerVariableExpn[u,m+1,x]},
    1/lst[[2]]*Subst[Int[NormalizeIntegrand[Simplify[lst[[1]]/x],x],x],x,(lst[[3]]*x)^lst[[2]] /;
    Not[FalseQ[lst]] && NeQ[lst[[2]],m+1]] /;
    IntegerQ[m] && NeQ[m,-1] && NonsumQ[u] && (GtQ[m,0] || Not[AlgebraicFunctionQ[u,x]])]
```

3: $\int x^m F[x] \, dx$ when $m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $x^m F[x] = k \left(x^{1/k}\right)^{k(m+1)-1} F\left[\left(x^{1/k}\right)^k\right] \partial_x x^{1/k}$

Rule: If $m \in \mathbb{F}$, let $k = \text{Denominator}[m]$, then

$$\int x^m F[x] \, dx \rightarrow k \text{Subst}\left[\int x^{k(m+1)-1} F[x^k] \, dx, x, x^{1/k}\right]$$

Program code:

```
Int[x^m*u_,x_Symbol] :=
  With[{k=Denominator[m]},
    k*Subst[Int[x^(k*(m+1)-1)*ReplaceAll[u,x->x^k],x],x,x^(1/k)] /;
  FractionQ[m]
```

4. $\int F\left[\sqrt{a+bx+cx^2}, x\right] \, dx$

1: $\int F\left[\sqrt{a+bx+cx^2}, x\right] \, dx$ when $a > 0$

Reference: G&R 2.251.1 (Euler substitution #1)

Derivation: Integration by substitution

Basis: $F\left[\sqrt{a+bx+cx^2}, x\right] =$

$$2 \text{Subst}\left[\frac{c\sqrt{a}-bx+\sqrt{a}x^2}{(c-x^2)^2} F\left[\frac{c\sqrt{a}-bx+\sqrt{a}x^2}{c-x^2}, \frac{-b+2\sqrt{a}x}{c-x^2}\right], x, \frac{-\sqrt{a}+\sqrt{a+bx+cx^2}}{x}\right] \partial_x \frac{-\sqrt{a}+\sqrt{a+bx+cx^2}}{x}$$

Rule: If $a > 0$, then

$$\int F\left[\sqrt{a + b x + c x^2}, x\right] dx \rightarrow 2 \operatorname{Subst}\left[\int \frac{c \sqrt{a} - b x + \sqrt{a} x^2}{(c - x^2)^2} F\left[\frac{c \sqrt{a} - b x + \sqrt{a} x^2}{c - x^2}, \frac{-b + 2 \sqrt{a} x}{c - x^2}\right] dx, x, \frac{-\sqrt{a} + \sqrt{a + b x + c x^2}}{x}\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_, x_Symbol] :=
  With[{lst=FunctionOfSquareRootOfQuadratic[u,x]},
    ShowStep["", "Int[F[Sqrt[a+b*x+c*x^2],x],x]",
      "2*Subst[Int[F[(c*Sqrt[a]-b*x+Sqrt[a]*x^2)/(c-x^2), (-b+2*Sqrt[a]*x)/(c-x^2)]*
        (c*Sqrt[a]-b*x+Sqrt[a]*x^2)/(c-x^2)^2,x],x, (-Sqrt[a]+Sqrt[a+b*x+c*x^2])/x]",
      Hold[2*Subst[Int[lst[[1]],x],x,lst[[2]]]]] /;
    Not[FalseQ[lst]] && EqQ[lst[[3]],1] /;
  SimplifyFlag && EulerIntegrandQ[u,x],

Int[u_, x_Symbol] :=
  With[{lst=FunctionOfSquareRootOfQuadratic[u,x]},
    2*Subst[Int[lst[[1]],x],x,lst[[2]]] /;
    Not[FalseQ[lst]] && EqQ[lst[[3]],1] /;
  EulerIntegrandQ[u,x]
```

2: $\int F\left[\sqrt{a + b x + c x^2}, x\right] dx$ when $a \neq 0 \wedge c > 0$

Reference: G&R 2.251.2 (Euler substitution #2)

Derivation: Integration by substitution

Basis: $F\left[\sqrt{a + b x + c x^2}, x\right] =$

$$2 \operatorname{Subst}\left[\frac{a \sqrt{c} + b x + \sqrt{c} x^2}{(b + 2 \sqrt{c} x)^2} F\left[\frac{a \sqrt{c} + b x + \sqrt{c} x^2}{b + 2 \sqrt{c} x}, \frac{-a + x^2}{b + 2 \sqrt{c} x}\right], x, \sqrt{c} x + \sqrt{a + b x + c x^2}\right]$$

$$\partial_x \left(\sqrt{c} x + \sqrt{a + b x + c x^2} \right)$$

Rule: If $a \neq 0 \wedge c > 0$, then

$$\int F[\sqrt{a + b x + c x^2}, x] dx \rightarrow 2 \operatorname{Subst}\left[\int \frac{a \sqrt{c} + b x + \sqrt{c} x^2}{(b + 2 \sqrt{c} x)^2} F\left[\frac{a \sqrt{c} + b x + \sqrt{c} x^2}{b + 2 \sqrt{c} x}, \frac{-a + x^2}{b + 2 \sqrt{c} x}\right] dx, x, \sqrt{c} x + \sqrt{a + b x + c x^2}\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_, x_Symbol] :=
  With[{lst=FunctionOfSquareRootOfQuadratic[u,x]},
    ShowStep["", "Int[F[Sqrt[a+b*x+c*x^2], x], x]",
      "2*Subst[Int[F[(a*Sqrt[c]+b*x+Sqrt[c]*x^2)/(b+2*Sqrt[c]*x), (-a+x^2)/(b+2*Sqrt[c]*x)]*
        (a*Sqrt[c]+b*x+Sqrt[c]*x^2)/(b+2*Sqrt[c]*x)^2, x], x, Sqrt[c]*x+Sqrt[a+b*x+c*x^2]]",
      Hold[2*Subst[Int[lst[[1]], x], x, lst[[2]]]]] /;
    Not[FalseQ[lst]] && EqQ[lst[[3]], 2] /;
  SimplifyFlag && EulerIntegrandQ[u,x],

Int[u_, x_Symbol] :=
  With[{lst=FunctionOfSquareRootOfQuadratic[u,x]},
    2*Subst[Int[lst[[1]], x], x, lst[[2]]] /;
    Not[FalseQ[lst]] && EqQ[lst[[3]], 2] /;
  EulerIntegrandQ[u,x]
```

3: $\int F[\sqrt{a + b x + c x^2}, x] dx$ when $a \neq 0 \wedge c \neq 0$

Reference: G&R 2.251.3 (Euler substitution #3)

Derivation: Integration by substitution

Basis:

$$F\left[\sqrt{a+bx+cx^2}, x\right] = -2\sqrt{b^2-4ac}$$

$$\text{Subst}\left[\frac{x}{(c-x^2)^2} F\left[-\frac{\sqrt{b^2-4ac}x}{c-x^2}, -\frac{bc+c\sqrt{b^2-4ac}+(-b+\sqrt{b^2-4ac})x^2}{2c(c-x^2)}\right], x, \frac{2c\sqrt{a+bx+cx^2}}{b-\sqrt{b^2-4ac}+2cx}\right] \partial_x \frac{2c\sqrt{a+bx+cx^2}}{b-\sqrt{b^2-4ac}+2cx}$$

Rule: If $a \neq 0 \wedge c \neq 0$, then

$$\int F\left[\sqrt{a+bx+cx^2}, x\right] dx \rightarrow -2\sqrt{b^2-4ac} \text{Subst}\left[\int \frac{x}{(c-x^2)^2} F\left[-\frac{\sqrt{b^2-4ac}x}{c-x^2}, -\frac{bc+c\sqrt{b^2-4ac}+(-b+\sqrt{b^2-4ac})x^2}{2c(c-x^2)}\right] dx, x, \frac{2c\sqrt{a+bx+cx^2}}{b-\sqrt{b^2-4ac}+2cx}\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfSquareRootOfQuadratic[u,x]},
    ShowStep["", "Int[F[Sqrt[a+b*x+c*x^2],x],x]",
      "-2*Sqrt[b^2-4*a*c]*Subst[Int[F[-Sqrt[b^2-4*a*c]*x/(c-x^2),
        (b*c+c*Sqrt[b^2-4*a*c]+(-b+Sqrt[b^2-4*a*c])*x^2]/(-2*c*(c-x^2))]*x/(c-x^2)^2,x],
        x,2*c*Sqrt[a+b*x+c*x^2]/(b-Sqrt[b^2-4*a*c]+2*c*x)]",
      Hold[2*Subst[Int[lst[[1]],x],x,lst[[2]]]]] /;
  Not[FalseQ[lst]] && EqQ[lst[[3]],3] /;
SimplifyFlag && EulerIntegrandQ[u,x],

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfSquareRootOfQuadratic[u,x]},
    2*Subst[Int[lst[[1]],x],x,lst[[2]]] /;
  Not[FalseQ[lst]] && EqQ[lst[[3]],3] /;
EulerIntegrandQ[u,x]
```

Algebraic expansion integration rules

$$1. \int \frac{1}{a + b v^n} dx \text{ when } n \in \mathbb{Z} \wedge n > 1$$

$$1. \int \frac{1}{a + b v^n} dx \text{ when } \frac{n}{2} \in \mathbb{Z}^+$$

$$\text{1: } \int \frac{1}{a + b v^2} dx$$

Derivation: Algebraic expansion

$$\text{Basis: } \frac{1}{a+b v^2} == \frac{1}{2 a \left(1 - \frac{v}{\sqrt{-a/b}}\right)} + \frac{1}{2 a \left(1 + \frac{v}{\sqrt{-a/b}}\right)}$$

Rule:

$$\int \frac{1}{a + b v^2} dx \rightarrow \frac{1}{2 a} \int \frac{1}{1 - \frac{v}{\sqrt{-a/b}}} dx + \frac{1}{2 a} \int \frac{1}{1 + \frac{v}{\sqrt{-a/b}}} dx$$

Program code:

```
Int[1/(a_+b_.*v_^2),x_Symbol] :=
(*1/(2*a)*Int[Together[1/(1-Rt[-b/a,2]*v)],x] + 1/(2*a)*Int[Together[1/(1+Rt[-b/a,2]*v)],x] /; *)
1/(2*a)*Int[Together[1/(1-v/Rt[-a/b,2])],x] + 1/(2*a)*Int[Together[1/(1+v/Rt[-a/b,2])],x] /;
FreeQ[{a,b},x]
```

2: $\int \frac{1}{a + b v^n} dx$ when $\frac{n}{2} \in \mathbb{Z} \wedge n > 2$

Derivation: Algebraic expansion

Basis: If $\frac{n}{2} \in \mathbb{Z}^+$, then $\frac{1}{a + b z^n} = \frac{2}{a n} \sum_{k=1}^{n/2} \frac{1}{1 - (-1)^{-4k/n} \left(-\frac{a}{b}\right)^{-2/n} z^2}$

Rule: If $\frac{n}{2} \in \mathbb{Z} \wedge n > 2$, then

$$\int \frac{1}{a + b v^n} dx \rightarrow \frac{2}{a n} \sum_{k=1}^{n/2} \int \frac{1}{1 - (-1)^{-4k/n} \left(-\frac{a}{b}\right)^{-2/n} v^2} dx$$

Program code:

```
Int[1/(a_+b_.*v_^n_),x_Symbol] :=
  Dist[2/(a*n),Sum[Int[Together[1/(1-v^2/((-1)^(4*k/n)*Rt[-a/b,n/2]))],x],{k,1,n/2}],x] /;
FreeQ[{a,b},x] && IGtQ[n/2,1]
```


2: $\int \frac{1}{a + b v^n} dx$ when $\frac{n-1}{2} \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: If $n \in \mathbb{Z}^+$, then $a + b z^n = a \prod_{k=1}^n \left(1 - (-1)^{-2k/n} \left(-\frac{a}{b} \right)^{-1/n} z \right)$

Basis: If $n \in \mathbb{Z}^+$, then $\frac{1}{a + b z^n} = \frac{1}{a n} \sum_{k=1}^n \frac{1}{1 - (-1)^{-2k/n} \left(-\frac{a}{b} \right)^{-1/n} z}$

Rule: If $\frac{n-1}{2} \in \mathbb{Z}^+$, then

$$\int \frac{1}{a + b v^n} dx \rightarrow \frac{1}{a n} \sum_{k=1}^n \int \frac{1}{1 - (-1)^{-2k/n} \left(-\frac{a}{b} \right)^{-1/n} v} dx$$

Program code:

```
Int[1/(a_+b_.*v_^n_),x_Symbol] :=
  Dist[1/(a*n),Sum[Int[Together[1/(1-v/((-1)^(2*k/n)*Rt[-a/b,n]))],x],{k,1,n}],x] /;
FreeQ[{a,b},x] && IGtQ[(n-1)/2,0]
```

2: $\int \frac{P_u}{a + b u^n} dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \frac{P_u}{a + b u^n} dx \rightarrow \int \left(\text{ExpandIntegrand} \left[\frac{P_x}{a + b x^n}, x \right] /. x \rightarrow u \right) dx$$

Program code:

```
Int[v_/(a_+b_.*u_^n_),x_Symbol] :=
  Int[ReplaceAll[ExpandIntegrand[PolynomialInSubst[v,u,x]/(a+b*x^n),x],x->u],x] /;
  FreeQ[{a,b},x] && IGtQ[n,0] && PolynomialInQ[v,u,x]
```

3: $\int u dx$ when $\text{NormalizeIntegrand}[u, x] \neq u$

Derivation: Algebraic simplification

Rule: If $\text{NormalizeIntegrand}[u, x] \neq u$, then

$$\int u dx \rightarrow \int \text{NormalizeIntegrand}[u, x] dx$$

Program code:

```
Int[u_,x_Symbol] :=
  With[{v=NormalizeIntegrand[u,x]},
    Int[v,x] /;
    v!=u]
```

4: $\int u \, dx$ when `ExpandIntegrand[u, x]` is a sum

Derivation: Algebraic expansion

Rule: If `ExpandIntegrand[u, x]` is a sum, then

$$\int u \, dx \rightarrow \int \text{ExpandIntegrand}[u, x] \, dx$$

Program code:

```
Int[u_, x_Symbol] :=
  With[{v = ExpandIntegrand[u, x]},
    Int[v, x] /;
    SumQ[v]]
```

Piecewise constant extraction integration rules

$$\mathbf{1:} \int u (a + b x^m)^p (c + d x^n)^q dx \text{ when } a + d == 0 \wedge b + c == 0 \wedge m + n == 0 \wedge p + q == 0$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(a + b x^m)^p}{x^{m p} \left(-b - \frac{a}{x^m}\right)^p} == 0$$

$$\text{Rule: If } a + d == 0 \wedge b + c == 0 \wedge m + n == 0 \wedge p + q == 0$$

$$\int u (a + b x^m)^p (c + d x^n)^q dx \rightarrow \frac{(a + b x^m)^p (c + d x^n)^q}{x^{m p}} \int u x^{m p} dx$$

Program code:

```
Int[u_.*(a_.+b_.*x_^m_.)^p_.*(c_.+d_.*x_^n_.)^q_., x_Symbol] :=
  (a+b*x^m)^p*(c+d*x^n)^q/x^(m*p)*Int[u*x^(m*p),x] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && EqQ[a+d,0] && EqQ[b+c,0] && EqQ[m+n,0] && EqQ[p+q,0]
```

$$\mathbf{2:} \int u (a + b x^n + c x^{2n})^p dx \text{ when } b^2 - 4 a c == 0 \wedge p + \frac{1}{2} \in \mathbb{Z}$$

Derivation: Piecewise constant extraction

$$\text{Basis: If } b^2 - 4 a c == 0 \wedge p - \frac{1}{2} \in \mathbb{Z}, \text{ then } (a + b x^n + c x^{2n})^p == \frac{\sqrt{a + b x^n + c x^{2n}}}{(4 c)^{p - \frac{1}{2}} (b + 2 c x^n)^2} (b + 2 c x^n)^{2 p}$$

$$\text{Basis: If } b^2 - 4 a c == 0, \text{ then } \partial_x \frac{\sqrt{a + b x^n + c x^{2n}}}{b + 2 c x^n} == 0$$

$$\text{Rule: If } b^2 - 4 a c == 0 \wedge p - \frac{1}{2} \in \mathbb{Z}, \text{ then}$$

$$\int u (a + b x^n + c x^{2n})^p dx \rightarrow \frac{\sqrt{a + b x^n + c x^{2n}}}{(4c)^{p-\frac{1}{2}} (b + 2cx^n)} \int u (b + 2cx^n)^{2p} dx$$

Program code:

```
Int[u*(a+b_.**x_^n_.+c_.**x_^n2_.)^p_, x_Symbol] :=
  Sqrt[a+b*x^n+c*x^(2*n)]/((4*c)^(p-1/2)*(b+2*c*x^n))*Int[u*(b+2*c*x^n)^(2*p), x] /;
FreeQ[{a,b,c,n,p}, x] && EqQ[n2, 2*n] && EqQ[b^2-4*a*c, 0] && IntegerQ[p-1/2]
```

Substitution integration rules

1: $\int F[(a + b x)^{1/n}, x] dx$ when $n \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $F[(a + b x)^{1/n}, x] = \frac{n}{b} \text{Subst}\left[x^{n-1} F\left[x, -\frac{a}{b} + \frac{x^n}{b}\right], x, (a + b x)^{1/n}\right] \partial_x (a + b x)^{1/n}$

Rule: If $n \in \mathbb{Z}$, then

$$\int F[(a + b x)^{1/n}, x] dx \rightarrow \frac{n}{b} \text{Subst}\left[\int x^{n-1} F\left[x, -\frac{a}{b} + \frac{x^n}{b}\right] dx, x, (a + b x)^{1/n}\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfLinear[u,x]},
    ShowStep["", "Int[F[(a+b*x)^(1/n),x],x]",
      "n/b*Subst[Int[x^(n-1)*F[x,-a/b+x^n/b],x],x,(a+b*x)^(1/n)"],Hold[
      lst[[2]]*lst[[4]]*Subst[Int[lst[[1]],x],x,lst[[3]]^(1/lst[[2]])]]] /;
    Not[FalseQ[lst]]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfLinear[u,x]},
    lst[[2]]*lst[[4]]*Subst[Int[lst[[1]],x],x,lst[[3]]^(1/lst[[2]])] /;
    Not[FalseQ[lst]]]
```

C: $\int u \, dx$

Rule:

$$\int u \, dx \rightarrow \int u \, dx$$

Program code:

```
Int[u_,x_] := CannotIntegrate[u,x]
```