Rules for integrands of the form $(d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p$

0. $\int (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p \, dx$ when $b^2 - 4 a c == 0$

   x: $\int (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p \, dx$ when $b^2 - 4 a c == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2 - 4 a c == 0$, then $a + b z + c z^2 == \frac{1}{c} \left(\frac{b}{2} + c z\right)^2$

Rule 1.2.3.2.4.1: If $b^2 - 4 a c == 0 \land p \in \mathbb{Z}$, then

$$\int (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p \, dx \longrightarrow \frac{1}{c^p} \int (d + e x^n)^q \left(\frac{b}{2} + c x^n\right)^{2p} dx$$

Program code:

```
(* Int[(d_+e_.*x_^n_.)^q_.*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/c^p*Int[(d+e*x^n)^q*(b/2+c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

2. $\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac=0 \land p \notin \mathbb{Z}$

1: $\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac=0 \land p \notin \mathbb{Z} \land 2cd-be=0$ **Necessary?**

Derivation: Piecewise constant extraction

Basis: If $b^2-4ac=0 \land 2cd-be=0$, then $\partial_x \frac{(a+b x^n+c x^{2n})^p}{(d+e x^n)^{2p}}=0$

Note: If $b^2-4ac=0 \land 2cd-be=0$, then $a+bz+cz^2=\frac{c}{e^2}(d+ez)^2$

Rule 1.2.3.3.0.1: If $b^2-4ac=0 \land p \notin \mathbb{Z} \land 2cd-be=0$, then

$$\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx \to \frac{(a+b x^n+c x^{2n})^p}{(d+e x^n)^{2p}} \int (d+e x^n)^{q+2p} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_.)^q_.*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^p/(d+e*x^n)^(2*p)*Int[(d+e*x^n)^(q+2*p),x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[2*c*d-b*e,0]
```

2: $\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac=0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4ac=0$, then $\partial_x \frac{(a+b x^n+c x^{2n})^p}{\left(\frac{b}{2}+c x^n\right)^{2p}}=0$

Note: If $b^2-4ac=0$, then $a+bz+cz^2=\frac{1}{c}\left(\frac{b}{2}+cz\right)^2$

Rule 1.2.3.3.0.2: If $b^2-4ac=0 \land p \notin \mathbb{Z}$, then

$$\int (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \frac{\left(a+b x^n+c x^{2n}\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]} \left(\frac{b}{2}+c x^n\right)^{2\,\text{FracPart}[p]}} \int (d+e x^n)^q \left(\frac{b}{2}+c x^n\right)^{2p} dx$$

Program code:

```
Int[(d_+e_.*x_^n_.)^q_.*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/(c^IntPart[p]*(b/2+c*x^n)^(2*FracPart[p]))*Int[(d+e*x^n)^q*(b/2+c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

1: $\int (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx$ when $(p \mid q) \in \mathbb{Z} \land n<0$

Derivation: Algebraic expansion

Basis: If $(p \mid q) \in \mathbb{Z}$, then $(d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p == x^{n(2p+q)} (e+d x^{-n})^q \left(c+b x^{-n}+a x^{-2n}\right)^p$

Rule 1.2.3.3.1: If $(p \mid q) \in \mathbb{Z} \land n<0$, then

$$\int (d+e x^n)^q \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \int x^{n(2p+q)} (e+d x^{-n})^q \left(c+b x^{-n}+a x^{-2n}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(n*(2*p+q))*(e+d*x^(-n))^q*(c+b*x^(-n)+a*x^(-2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && IntegersQ[p,q] && NegQ[n]
```

```
Int[(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(n*(2*p+q))*(e+d*x^(-n))^q*(c+a*x^(-2*n))^p,x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && IntegersQ[p,q] && NegQ[n]
```

2: $\int (d + e\,x^n)^q\,(a + b\,x^n + c\,x^{2\,n})^p\,dx$ when $n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $F[x^n] = -\text{Subst}\left[\frac{F[x^{-n}]}{x^2},\, x,\, \frac{1}{x}\right]\,\partial_x \frac{1}{x}$

Rule 1.2.3.3.2: If $n \in \mathbb{Z}^-$, then

$$\int (d + e\,x^n)^q\,(a + b\,x^n + c\,x^{2\,n})^p\,dx \longrightarrow -\text{Subst}\left[\int \frac{(d + e\,x^{-n})^q\,(a + b\,x^{-n} + c\,x^{-2\,n})^p}{x^2}\,dx,\, x,\, \frac{1}{x}\right]$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(a_.+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  -Subst[Int[(d+e*x^(-n))^q*(a+b*x^(-n)+c*x^(-2*n))^p/x^2,x],x,1/x] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[n2,2*n] && ILtQ[n,0]
```

```
Int[(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  -Subst[Int[(d+e*x^(-n))^q*(a+c*x^(-2*n))^p/x^2,x],x,1/x] /;
FreeQ[{a,c,d,e,p,q},x] && EqQ[n2,2*n] && ILtQ[n,0]
```

3: $\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g\in\mathbb{Z}^+$, then $F[x^n] = g\,\text{Subst}\left[x^{g-1}\,F[x^{g\,n}],\,x,\,x^{1/g}\right]\,\partial_x x^{1/g}$

Rule 1.2.3.3.3: If $n\in\mathbb{F}$, let $g = \text{Denominator}[n]$, then

$$\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx\;\longrightarrow\;g\,\text{Subst}\left[\int x^{g-1}\,(d+e\,x^{g\,n})^q\,(a+b\,x^{g\,n}+c\,x^{2\,g\,n})^p\,dx,\,x,\,x^{1/g}\right]$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(a_.+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g-1)*(d+e*x^(g*n))^q*(a+b*x^(g*n)+c*x^(2*g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[n2,2*n] && FractionQ[n]
```

```
Int[(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g-1)*(d+e*x^(g*n))^q*(a+c*x^(2*g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,c,d,e,p,q},x] && EqQ[n2,2*n] && FractionQ[n]
```

4. $\int (d + e x^n)^q (b x^n + c x^{2n})^p \, dx$ when $p \notin \mathbb{Z}$

1. $\int (d + e x^n) (b x^n + c x^{2n})^p \, dx$ when $p \notin \mathbb{Z}$

1: $\int (d + e x^n) (b x^n + c x^{2n})^p \, dx$ when $p \notin \mathbb{Z} \land n (2p + 1) + 1 == 0$

Derivation: Trinomial recurrence 2a with $a = 0, m = 0$ and $n (2p + 1) + 1 == 0$ composed with trinomial recurrene 5 with $a = 0$

Rule 1.2.3.3.4.1.1: If $p \notin \mathbb{Z} \land n (2p + 1) + 1 == 0$, then

$$\int (d + e x^n) (b x^n + c x^{2n})^p \, dx \longrightarrow -\frac{(c d - b e) (b x^n + c x^{2n})^{p+1}}{b c n (p + 1) x^{2n(p+1)}} + \frac{e}{c} \int x^{-n} (b x^n + c x^{2n})^{p+1} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)*(b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  (b*e-d*c)*(b*x^n+c*x^(2*n))^(p+1)/(b*c*n*(p+1)*x^(2*n*(p+1))) +
  e/c*Int[x^(-n)*(b*x^n+c*x^(2*n))^(p+1),x] /;
FreeQ[{b,c,d,e,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[p]] && EqQ[n*(2*p+1)+1,0]
```

2: $\int (d+e x^n)\,(b x^n+c x^{2n})^p\,dx$ when $p\notin\mathbb{Z}\land n(2p+1)+1\neq 0\land b e(np+1)-c d(n(2p+1)+1)==0$

Derivation: Trinomial recurrence 3a with $a=0$ with $b e(np+1)-c d(n(2p+1)+1)==0$

Rule 1.2.3.3.4.1.2: If $p\notin\mathbb{Z}\land n(2p+1)+1\neq 0\land b e(np+1)-c d(n(2p+1)+1)==0$, then

$$\int (d+e x^n)\,(b x^n+c x^{2n})^p\,dx\;\longrightarrow\;\frac{e x^{-n+1}\,(b x^n+c x^{2n})^{p+1}}{c\,(n(2p+1)+1)}$$

Program code:

```
Int[(d_+e_.*x_^n_)*(b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  e*x^(-n+1)*(b*x^n+c*x^(2*n))^(p+1)/(c*(n*(2*p+1)+1)) /;
FreeQ[{b,c,d,e,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[p]] && NeQ[n*(2*p+1)+1,0] && EqQ[b*e*(n*p+1)-c*d*(n*(2*p+1)+1),0]
```

3: $\int (d+e x^n)\,(b x^n+c x^{2n})^p\,dx$ when $p\notin\mathbb{Z}\land n(2p+1)+1\neq 0\land b e(np+1)-c d(n(2p+1)+1)\neq 0$

Derivation: Trinomial recurrence 3a with $a=0$

Rule 1.2.3.3.4.1.3: If $p\notin\mathbb{Z}\land n(2p+1)+1\neq 0\land b e(np+1)-c d(n(2p+1)+1)\neq 0$, then

$$\int (d+e x^n)\,(b x^n+c x^{2n})^p\,dx\;\longrightarrow$$
$$\frac{e x^{-n+1}\,(b x^n+c x^{2n})^{p+1}}{c\,(n(2p+1)+1)}-\frac{b e(np+1)-c d(n(2p+1)+1)}{c\,(n(2p+1)+1)}\int (b x^n+c x^{2n})^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)*(b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  e*x^(-n+1)*(b*x^n+c*x^(2*n))^(p+1)/(c*(n*(2*p+1)+1)) -
  (b*e*(n*p+1)-c*d*(n*(2*p+1)+1))/(c*(n*(2*p+1)+1))*Int[(b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{b,c,d,e,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[p]] && NeQ[n*(2*p+1)+1,0] && NeQ[b*e*(n*p+1)-c*d*(n*(2*p+1)+1),0]
```

2: $\int (d+e\,x^n)^q\,(b\,x^n+c\,x^{2\,n})^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,x^n+c\,x^{2\,n})^p}{x^{n\,p}\,(b+c\,x^n)^p}==0$

Basis: $\frac{(b\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}{x^{n\,\text{FracPart}[p]}\,(b+c\,x^n)^{\text{FracPart}[p]}}==\frac{(b\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}{x^{n\,\text{FracPart}[p]}\,(b+c\,x^n)^{\text{FracPart}[p]}}$

Rule 1.2.3.3.4.2: If $p\notin\mathbb{Z}$, then

$$\int (d+e\,x^n)^q\,(b\,x^n+c\,x^{2\,n})^p\,dx\,\rightarrow\,\frac{(b\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}{x^{n\,\text{FracPart}[p]}\,(b+c\,x^n)^{\text{FracPart}[p]}}\int x^{n\,p}\,(d+e\,x^n)^q\,(b+c\,x^n)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  (b*x^n+c*x^(2*n))^FracPart[p]/(x^(n*FracPart[p])*(b+c*x^n)^FracPart[p])*Int[x^(n*p)*(d+e*x^n)^q*(b+c*x^n)^p,x] /;
FreeQ[{b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && Not[IntegerQ[p]]
```

6. $\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 == 0$

1: $\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c d^2-b d e+a e^2 == 0$, then $a+b z+c z^2 == (d+e z)\left(\frac{a}{d}+\frac{c z}{e}\right)$

Rule 1.2.3.3.6.1: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \in \mathbb{Z}$, then

$$\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx \longrightarrow \int (d+e x^n)^{p+q} \left(\frac{a}{d}+\frac{c x^n}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(d+e*x^n)^(p+q)*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,n,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(d+e*x^n)^(p+q)*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,c,d,e,n,q},x] && EqQ[n2,2*n] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

2: $\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c\,d^2-b\,d\,e+a\,e^2=0$, then $\partial_x\,\frac{(a+b\,x^n+c\,x^{2\,n})^p}{(d+e\,x^n)^p\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^p}=0$

Basis: If $c\,d^2-b\,d\,e+a\,e^2=0$, then $\frac{(a+b\,x^n+c\,x^{2\,n})^p}{(d+e\,x^n)^p\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^p}=\frac{(a+b\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}{(d+e\,x^n)^{\text{FracPart}[p]}\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^{\text{FracPart}[p]}}$

Rule 1.2.3.3.6.2: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,p\notin\mathbb{Z}$, then

$$\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx\,\rightarrow\,\frac{(a+b\,x^n+c\,x^{2\,n})^{\text{FracPart}[p]}}{(d+e\,x^n)^{\text{FracPart}[p]}\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^{\text{FracPart}[p]}}\int (d+e\,x^n)^{p+q}\,\left(\frac{a}{d}+\frac{c\,x^n}{e}\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/((d+e*x^n)^FracPart[p]*(a/d+c*x^n/e)^FracPart[p])*Int[(d+e*x^n)^(p+q)*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]]
```

```
Int[(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  (a+c*x^(2*n))^FracPart[p]/((d+e*x^n)^FracPart[p]*(a/d+c*x^n/e)^FracPart[p])*Int[(d+e*x^n)^(p+q)*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,c,d,e,n,p,q},x] && EqQ[n2,2*n] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]]
```

7. $\int (d + e x^n)^q (a + b x^n + c x^{2n}) \, dx$ when $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0$

1: $\int (d + e x^n)^q (a + b x^n + c x^{2n}) \, dx$ when $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0 \wedge q \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.3.3.7.1: If $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0 \wedge q \in \mathbb{Z}^+$, then

$$\int (d + e x^n)^q (a + b x^n + c x^{2n}) \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(d + e x^n)^q (a + b x^n + c x^{2n}), x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q*(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[q,0]
```

```
Int[(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q*(a+c*x^(2*n)),x],x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && IGtQ[q,0]
```

2: $\int (d+e x^n)^q \left(a+b x^n+c x^{2n}\right) dx$ when $b^2-4ac\neq 0 \wedge c d^2-b d e+a e^2\neq 0 \wedge q<-1$

Derivation: ???

Rule 1.2.3.3.7.2: If $b^2-4ac\neq 0 \wedge c d^2-b d e+a e^2\neq 0 \wedge q<-1$, then

$$\int (d+e x^n)^q \left(a+b x^n+c x^{2n}\right) dx \longrightarrow$$

$$-\frac{\left(c d^2-b d e+a e^2\right) x \left(d+e x^n\right)^{q+1}}{d e^2 n (q+1)}+\frac{1}{n (q+1) d e^2}\int \left(d+e x^n\right)^{q+1} \left(c d^2-b d e+a e^2 (n (q+1)+1)+c d e n (q+1) x^n\right) dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  -(c*d^2-b*d*e+a*e^2)*x*(d+e*x^n)^(q+1)/(d*e^2*n*(q+1)) +
  1/(n*(q+1)*d*e^2)*Int[(d+e*x^n)^(q+1)*Simp[c*d^2-b*d*e+a*e^2*(n*(q+1)+1)+c*d*e*n*(q+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[q,-1]
```

```
Int[(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_),x_Symbol] :=
  -(c*d^2+a*e^2)*x*(d+e*x^n)^(q+1)/(d*e^2*n*(q+1)) +
  1/(n*(q+1)*d*e^2)*Int[(d+e*x^n)^(q+1)*Simp[c*d^2+a*e^2*(n*(q+1)+1)+c*d*e*n*(q+1)*x^n,x],x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && LtQ[q,-1]
```

3: $\int (d+e x^n)^q (a+b x^n+c x^{2n})\,dx$ when $b^2-4ac\neq 0\wedge c d^2-b d e+a e^2\neq 0$

Derivation: Special case of rule for $P_q[x]\,(d+e x^n)^q$

Rule 1.2.3.3.7.3: If $b^2-4ac\neq 0\wedge c d^2-b d e+a e^2\neq 0$, then

$$\int (d+e x^n)^q (a+b x^n+c x^{2n})\,dx\ \longrightarrow$$

$$\frac{c\,x^{n+1}\,(d+e x^n)^{q+1}}{e\,(n\,(q+2)+1)}+\frac{1}{e\,(n\,(q+2)+1)}\int (d+e x^n)^q \left(a e\,(n\,(q+2)+1)-(c d\,(n+1)-b e\,(n\,(q+2)+1))\,x^n\right)dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  c*x^(n+1)*(d+e*x^n)^(q+1)/(e*(n*(q+2)+1)) +
  1/(e*(n*(q+2)+1))*Int[(d+e*x^n)^q*(a*e*(n*(q+2)+1)-(c*d*(n+1)-b*e*(n*(q+2)+1))*x^n),x] /;
FreeQ[{a,b,c,d,e,n,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_),x_Symbol] :=
  c*x^(n+1)*(d+e*x^n)^(q+1)/(e*(n*(q+2)+1)) +
  1/(e*(n*(q+2)+1))*Int[(d+e*x^n)^q*(a*e*(n*(q+2)+1)-c*d*(n+1)*x^n),x] /;
FreeQ[{a,c,d,e,n,q},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0]
```

8. $\int \frac{(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

1. $\int \frac{(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,q\in\mathbb{Z}$

1. $\int \frac{d+e\,x^n}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0$

1. $\int \frac{d+e\,x^n}{a+c\,x^{2\,n}}\,dx$ when $c\,d^2+a\,e^2\neq 0$

1. $\int \frac{d+e\,x^n}{a+c\,x^{2\,n}}\,dx$ when $c\,d^2+a\,e^2\neq 0\,\land\,c\,d^2-a\,e^2==0\,\land\,\frac{n}{2}\in\mathbb{Z}^+$

1: $\int \frac{d+e\,x^n}{a+c\,x^{2\,n}}\,dx$ when $c\,d^2-a\,e^2==0\,\land\,\frac{n}{2}\in\mathbb{Z}^+\,\land\,d\,e>0$

Derivation: Algebraic expansion

Basis: If $c\,d^2-a\,e^2==0$ and $q\to\sqrt{2\,d\,e}$, then $\frac{d+e\,z^2}{a+c\,z^4}==\frac{e^2}{2\,c\,(d+q\,z+e\,z^2)}+\frac{e^2}{2\,c\,(d-q\,z+e\,z^2)}$

Rule 1.2.3.3.8.1.1.1.1.1: If $c\,d^2-a\,e^2==0\,\land\,\frac{n}{2}\in\mathbb{Z}^+\,\land\,d\,e>0$, let $q\to\sqrt{2\,d\,e}$, then

$$\int \frac{d+e\,x^n}{a+c\,x^{2\,n}}\,dx\;\longrightarrow\;\frac{e^2}{2\,c}\int\frac{1}{d+q\,x^{n/2}+e\,x^n}\,dx+\frac{e^2}{2\,c}\int\frac{1}{d-q\,x^{n/2}+e\,x^n}\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[2*d*e,2]},
  e^2/(2*c)*Int[1/(d+q*x^(n/2)+e*x^n),x] + e^2/(2*c)*Int[1/(d-q*x^(n/2)+e*x^n),x]] /;
FreeQ[{a,c,d,e},x] && EqQ[n2,2*n] && EqQ[c*d^2-a*e^2,0] && IGtQ[n/2,0] && PosQ[d*e]
```

2: $\int \frac{d+e\,x^n}{a+c\,x^{2\,n}}\,dx$ when $c\,d^2-a\,e^2==0\,\land\,\frac{n}{2}\in\mathbb{Z}^+\,\land\,d\,e\not>0$

Derivation: Algebraic expansion

Basis: If $c\,d^2 - a\,e^2 = 0$, let $q = \sqrt{-2\,d\,e}$ then $\frac{d+e\,z^2}{a+c\,z^4} = \frac{d\,(d-q\,z)}{2\,a\,(d-q\,z-e\,z^2)} + \frac{d\,(d+q\,z)}{2\,a\,(d+q\,z-e\,z^2)}$

Rule 1.2.3.3.8.1.1.1.1.2: If $c\,d^2 - a\,e^2 = 0 \land \frac{n}{2} \in \mathbb{Z}^+ \land d\,e \not> 0$, let $q \to \sqrt{-2\,d\,e}$, then

$$\int \frac{d+e\,x^n}{a+c\,x^{2\,n}}\,dx \longrightarrow \frac{d}{2\,a}\int \frac{d-q\,x^{n/2}}{d-q\,x^{n/2}-e\,x^n}\,dx + \frac{d}{2\,a}\int \frac{d+q\,x^{n/2}}{d+q\,x^{n/2}-e\,x^n}\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[-2*d*e,2]},
  d/(2*a)*Int[(d-q*x^(n/2))/(d-q*x^(n/2)-e*x^n),x] +
  d/(2*a)*Int[(d+q*x^(n/2))/(d+q*x^(n/2)-e*x^n),x]] /;
FreeQ[{a,c,d,e},x] && EqQ[n2,2*n] && EqQ[c*d^2-a*e^2,0] && IGtQ[n/2,0] && NegQ[d*e]
```

2: $\int \frac{d+e\,x^n}{a+c\,x^{2n}}\,dx$ when $c\,d^2+a\,e^2\neq 0\,\wedge\,c\,d^2-a\,e^2\neq 0\,\wedge\,\frac{n}{2}\in\mathbb{Z}^+\wedge\,a\,c>0$

Derivation: Algebraic expansion

Basis: If $q\to\left(\frac{a}{c}\right)^{1/4}$, then $\frac{d+e\,z^2}{a+c\,z^4}==\frac{\sqrt{2}\,d\,q-\left(d-e\,q^2\right)\,z}{2\,\sqrt{2}\,c\,q^3\,\left(q^2-\sqrt{2}\,q\,z+z^2\right)}+\frac{\sqrt{2}\,d\,q+\left(d-e\,q^2\right)\,z}{2\,\sqrt{2}\,c\,q^3\,\left(q^2+\sqrt{2}\,q\,z+z^2\right)}$

Rule 1.2.3.3.8.1.1.1.2.2: If $c\,d^2+a\,e^2\neq 0\,\wedge\,c\,d^2-a\,e^2\neq 0\,\wedge\,\frac{n}{2}\in\mathbb{Z}^+\wedge\,a\,c>0$, let $q\to\left(\frac{a}{c}\right)^{1/4}$, then

$$\int\frac{d+e\,x^n}{a+c\,x^{2n}}\,dx\,\longrightarrow\,\frac{1}{2\,\sqrt{2}\,c\,q^3}\int\frac{\sqrt{2}\,d\,q-\left(d-e\,q^2\right)\,x^{n/2}}{q^2-\sqrt{2}\,q\,x^{n/2}+x^n}\,dx+\frac{1}{2\,\sqrt{2}\,c\,q^3}\int\frac{\sqrt{2}\,d\,q+\left(d-e\,q^2\right)\,x^{n/2}}{q^2+\sqrt{2}\,q\,x^{n/2}+x^n}\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[a/c,4]},
  1/(2*Sqrt[2]*c*q^3)*Int[(Sqrt[2]*d*q-(d-e*q^2)*x^(n/2))/(q^2-Sqrt[2]*q*x^(n/2)+x^n),x] +
  1/(2*Sqrt[2]*c*q^3)*Int[(Sqrt[2]*d*q+(d-e*q^2)*x^(n/2))/(q^2+Sqrt[2]*q*x^(n/2)+x^n),x]] /;
FreeQ[{a,c,d,e},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && IGtQ[n/2,0] && PosQ[a*c]
```

3: $\int \frac{d+e x^3}{a+c x^6} \, dx$ when $c d^2+a e^2 \neq 0 \land \frac{c}{a}>0$

Derivation: Algebraic expansion

Basis: Let $q \to \left(\frac{c}{a}\right)^{1/6}$, then $\frac{d+e x^3}{a+c x^6} == \frac{q^2 d-e x}{3 a q^2 \left(1+q^2 x^2\right)}+\frac{2 q^2 d-\left(\sqrt{3} q^3 d-e\right) x}{6 a q^2 \left(1-\sqrt{3} q x+q^2 x^2\right)}+\frac{2 q^2 d+\left(\sqrt{3} q^3 d+e\right) x}{6 a q^2 \left(1+\sqrt{3} q x+q^2 x^2\right)}$

Rule 1.2.3.3.8.1.1.1.3: If $c d^2+a e^2 \neq 0 \land \frac{c}{a}>0$, let $q \to \left(\frac{c}{a}\right)^{1/6}$, then

$$\int \frac{d+e x^3}{a+c x^6} \, dx \to \frac{1}{3 a q^2} \int \frac{q^2 d-e x}{1+q^2 x^2} \, dx+\frac{1}{6 a q^2} \int \frac{2 q^2 d-\left(\sqrt{3} q^3 d-e\right) x}{1-\sqrt{3} q x+q^2 x^2} \, dx+\frac{1}{6 a q^2} \int \frac{2 q^2 d+\left(\sqrt{3} q^3 d+e\right) x}{1+\sqrt{3} q x+q^2 x^2} \, dx$$

Program code:

```
Int[(d_+e_.*x_^3)/(a_+c_.*x_^6),x_Symbol] :=
  With[{q=Rt[c/a,6]},
  1/(3*a*q^2)*Int[(q^2*d-e*x)/(1+q^2*x^2),x] +
  1/(6*a*q^2)*Int[(2*q^2*d-(Sqrt[3]*q^3*d-e)*x)/(1-Sqrt[3]*q*x+q^2*x^2),x] +
  1/(6*a*q^2)*Int[(2*q^2*d+(Sqrt[3]*q^3*d+e)*x)/(1+Sqrt[3]*q*x+q^2*x^2),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && PosQ[c/a]
```

4: $\int \frac{d+e x^n}{a+c x^{2n}} \, dx$ when $c d^2 + a e^2 \neq 0 \land a c \ngtr 0 \land n \in \mathbb{Z}$

Derivation: Algebraic expansion

- Basis: If $q \to \sqrt{-\frac{a}{c}}$, then $\frac{d+e z}{a+c z^2} == \frac{d+e q}{2 (a+c q z)} + \frac{d-e q}{2 (a-c q z)}$

- Rule 1.2.3.3.8.1.1.1.4: If $c d^2 + a e^2 \neq 0 \land a c \ngtr 0 \land n \in \mathbb{Z}$, let $q \to \sqrt{-\frac{a}{c}}$, then

$$\int \frac{d+e x^n}{a+c x^{2n}} \, dx \longrightarrow \frac{d+e q}{2} \int \frac{1}{a+c q x^n} \, dx + \frac{d-e q}{2} \int \frac{1}{a-c q x^n} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[-a/c,2]},
  (d+e*q)/2*Int[1/(a+c*q*x^n),x] + (d-e*q)/2*Int[1/(a-c*q*x^n),x]] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && NegQ[a*c] && IntegerQ[n]
```

5: $\int \frac{d+e x^n}{a+c x^{2n}} \, dx$ when $c d^2 + a e^2 \neq 0 \land (a c > 0 \lor n \notin \mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.3.3.8.1.1.1.5: If $c d^2 + a e^2 \neq 0 \land (a c > 0 \lor n \notin \mathbb{Z})$, then

$$\int \frac{d+e x^n}{a+c x^{2n}} \, dx \longrightarrow d \int \frac{1}{a+c x^{2n}} \, dx + e \int \frac{x^n}{a+c x^{2n}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+c_.*x_^n2_),x_Symbol] :=
  d*Int[1/(a+c*x^(2*n)),x] + e*Int[x^n/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && (PosQ[a*c] || Not[IntegerQ[n]])
```

2. $\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \wedge c d^2-b d e+a e^2 \neq 0$

1. $\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \wedge c d^2-a e^2 == 0 \wedge \frac{n}{2} \in \mathbb{Z}^+$

1: $\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \wedge c d^2-a e^2 == 0 \wedge \frac{n}{2} \in \mathbb{Z}^+ \wedge \frac{2d}{e}-\frac{b}{c} > 0$

Derivation: Algebraic expansion

Basis: If $c d^2-a e^2 == 0$ and $q \to \sqrt{\frac{2d}{e}-\frac{b}{c}}$, then $\frac{d+e z^2}{a+b z^2+c z^4} == \frac{e^2}{2c\,(d+e q z+e z^2)} + \frac{e^2}{2c\,(d-e q z+e z^2)}$

Rule 1.2.3.3.8.1.1.2.1.1: If $b^2-4ac \neq 0 \wedge c d^2-a e^2 == 0 \wedge \frac{n}{2} \in \mathbb{Z}^+ \wedge \frac{2d}{e}-\frac{b}{c} > 0$, let $q \to \sqrt{\frac{2d}{e}-\frac{b}{c}}$, then

$$\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx \to \frac{e}{2c} \int \frac{1}{\frac{d}{e}+q x^{n/2}+x^n} \, dx + \frac{e}{2c} \int \frac{1}{\frac{d}{e}-q x^{n/2}+x^n} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[2*d/e-b/c,2]},
  e/(2*c)*Int[1/Simp[d/e+q*x^(n/2)+x^n,x],x] +
  e/(2*c)*Int[1/Simp[d/e-q*x^(n/2)+x^n,x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-a*e^2,0] && IGtQ[n/2,0] && (GtQ[2*d/e-b/c,0] || Not[LtQ[2*d/e-b
```

2: $\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land \frac{n}{2} \in \mathbb{Z}^+ \land b^2-4 a c > 0$

Derivation: Algebraic expansion

- Basis: Let $q \to \sqrt{b^2-4 a c}$, then $\frac{d+e z}{a+b z+c z^2} == \left(\frac{e}{2}+\frac{2 c d-b e}{2 q}\right) \frac{1}{\frac{b}{2}-\frac{q}{2}+c z}+\left(\frac{e}{2}-\frac{2 c d-b e}{2 q}\right) \frac{1}{\frac{b}{2}+\frac{q}{2}+c z}$

Rule 1.2.3.3.8.1.1.2.1.2: If $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land \frac{n}{2} \in \mathbb{Z}^+ \land b^2-4 a c > 0$, let $q \to \sqrt{b^2-4 a c}$, then

$$\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx \longrightarrow \left(\frac{e}{2}+\frac{2 c d-b e}{2 q}\right) \int \frac{1}{\frac{b}{2}-\frac{q}{2}+c x^n} \, dx+\left(\frac{e}{2}-\frac{2 c d-b e}{2 q}\right) \int \frac{1}{\frac{b}{2}+\frac{q}{2}+c x^n} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (e/2+(2*c*d-b*e)/(2*q))*Int[1/(b/2-q/2+c*x^n),x] + (e/2-(2*c*d-b*e)/(2*q))*Int[1/(b/2+q/2+c*x^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-a*e^2,0] && IGtQ[n/2,0] && GtQ[b^2-4*a*c,0]
```

3: $\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \land c d^2-a e^2 == 0 \land \frac{n}{2} \in \mathbb{Z}^+ \land b^2-4ac \ngtr 0$

Derivation: Algebraic expansion

Basis: If $c d^2-a e^2 == 0$ and $q \to \sqrt{-\frac{2d}{e}-\frac{b}{c}}$, then $\frac{d+e z^2}{a+b z^2+c z^4} == \frac{e (q-2z)}{2cq \left(\frac{d}{e}+q z-z^2\right)} + \frac{e (q+2z)}{2cq \left(\frac{d}{e}-q z-z^2\right)}$

Rule 1.2.3.3.8.1.1.2.1.3: If $b^2-4ac \neq 0 \land c d^2-a e^2 == 0 \land \frac{n}{2} \in \mathbb{Z}^+ \land b^2-4ac \ngtr 0$, let $q \to \sqrt{-\frac{2d}{e}-\frac{b}{c}}$, then

$$\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx \to \frac{e}{2cq} \int \frac{q-2x^{n/2}}{\frac{d}{e}+q x^{n/2}-x^n} \, dx + \frac{e}{2cq} \int \frac{q+2x^{n/2}}{\frac{d}{e}-q x^{n/2}-x^n} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[-2*d/e-b/c,2]},
  e/(2*c*q)*Int[(q-2*x^(n/2))/Simp[d/e+q*x^(n/2)-x^n,x],x] +
  e/(2*c*q)*Int[(q+2*x^(n/2))/Simp[d/e-q*x^(n/2)-x^n,x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-a*e^2,0] && IGtQ[n/2,0] && Not[GtQ[b^2-4*a*c,0]]
```

2: $\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land \left(b^2-4ac > 0 \lor \frac{n}{2} \notin \mathbb{Z}^+\right)$

Derivation: Algebraic expansion

Basis: Let $q \to \sqrt{b^2-4ac}$, then $\frac{d+e z}{a+b z+c z^2} == \left(\frac{e}{2}+\frac{2cd-be}{2q}\right) \frac{1}{\frac{b}{2}-\frac{q}{2}+c z} + \left(\frac{e}{2}-\frac{2cd-be}{2q}\right) \frac{1}{\frac{b}{2}+\frac{q}{2}+c z}$

Rule 1.2.3.3.8.1.1.2.2: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land \left(b^2-4ac > 0 \lor \frac{n}{2} \notin \mathbb{Z}^+\right)$, let $q \to \sqrt{b^2-4ac}$, then

$$\int \frac{d + e\, x^n}{a + b\, x^n + c\, x^{2\,n}}\, dx \longrightarrow \left(\frac{e}{2} + \frac{2\,c\,d - b\,e}{2\,q}\right) \int \frac{1}{\frac{b}{2} - \frac{q}{2} + c\, x^n}\, dx + \left(\frac{e}{2} - \frac{2\,c\,d - b\,e}{2\,q}\right) \int \frac{1}{\frac{b}{2} + \frac{q}{2} + c\, x^n}\, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (e/2+(2*c*d-b*e)/(2*q))*Int[1/(b/2-q/2+c*x^n),x] + (e/2-(2*c*d-b*e)/(2*q))*Int[1/(b/2+q/2+c*x^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && (PosQ[b^2-4*a*c] || Not[IGtQ[n/2,0]])
```

3: $\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge \frac{n}{2} \in \mathbb{Z}^+ \wedge b^2-4ac \not> 0$

Derivation: Algebraic expansion

Basis: If $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2q-\frac{b}{c}}$, then $\frac{d+e z^2}{a+b z^2+c z^4} = \frac{d r-(d-e q) z}{2 c q r (q-r z+z^2)} + \frac{d r+(d-e q) z}{2 c q r (q+r z+z^2)}$

Note: If $(a \mid b \mid c) \in \mathbb{R} \wedge b^2-4ac<0$, then $\frac{a}{c}>0$ and $2\sqrt{\frac{a}{c}}-\frac{b}{c}>0$.

Rule 1.2.3.3.8.1.1.2.3: If $b^2-4ac \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge \frac{n}{2} \in \mathbb{Z}^+ \wedge b^2-4ac \not> 0$, let $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2q-\frac{b}{c}}$, then

$$\int \frac{d+e x^n}{a+b x^n+c x^{2n}} \, dx \longrightarrow \frac{1}{2 c q r} \int \frac{d r-(d-e q) x^{n/2}}{q-r x^{n/2}+x^n} \, dx + \frac{1}{2 c q r} \int \frac{d r+(d-e q) x^{n/2}}{q+r x^{n/2}+x^n} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*q*r)*Int[(d*r-(d-e*q)*x^(n/2))/(q-r*x^(n/2)+x^n),x] +
  1/(2*c*q*r)*Int[(d*r+(d-e*q)*x^(n/2))/(q+r*x^(n/2)+x^n),x]]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[n/2,0] && NegQ[b^2-4*a*c]
```

2: $\int \frac{(d+e x^n)^q}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge q \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.3.3.8.1.2: If $b^2-4ac \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge q \in \mathbb{Z}$, then

$$\int \frac{(d+e x^n)^q}{a+b x^n+c x^{2n}} \, dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{(d+e x^n)^q}{a+b x^n+c x^{2n}}, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[q]
```

```
Int[(d_+e_.*x_^n_)^q_/(a_+c_.*x_^n2_),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q/(a+c*x^(2*n)),x],x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && IntegerQ[q]
```

2. $\int \frac{(d+e x^n)^q}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z}$

1: $\int \frac{(d+e x^n)^q}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z} \land q < -1$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b z+c z^2} == \frac{e^2}{c d^2-b d e+a e^2} + \frac{(d+e z)(c d-b e-c e z)}{(c d^2-b d e+a e^2)(a+b z+c z^2)}$

Rule 1.2.3.3.8.2.1: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z} \land q < -1$, then

$$\int \frac{(d+e x^n)^q}{a+b x^n+c x^{2n}} \, dx \longrightarrow \frac{e^2}{c d^2-b d e+a e^2} \int (d+e x^n)^q \, dx + \frac{1}{c d^2-b d e+a e^2} \int \frac{(d+e x^n)^{q+1}(c d-b e-c e x^n)}{a+b x^n+c x^{2n}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  e^2/(c*d^2-b*d*e+a*e^2)*Int[(d+e*x^n)^q,x] +
  1/(c*d^2-b*d*e+a*e^2)*Int[(d+e*x^n)^(q+1)*(c*d-b*e-c*e*x^n)/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[q]] && LtQ[q,-1]
```

```
Int[(d_+e_.*x_^n_)^q_/(a_+c_.*x_^n2_),x_Symbol] :=
  e^2/(c*d^2+a*e^2)*Int[(d+e*x^n)^q,x] +
  c/(c*d^2+a*e^2)*Int[(d+e*x^n)^(q+1)*(d-e*x^n)/(a+c*x^(2*n)),x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[q]] && LtQ[q,-1]
```

2: $\int \frac{(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ q\notin\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $r=\sqrt{b^2-4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2}==\frac{2\,c}{r\,(b-r+2\,c\,z)}-\frac{2\,c}{r\,(b+r+2\,c\,z)}$

Rule 1.2.3.3.8.2.2: If $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ q\notin\mathbb{Z}$, then

$$\int \frac{(d+e\,x^n)^q}{a+b\,x^n+c\,x^{2\,n}}\,dx\ \longrightarrow\ \frac{2\,c}{r}\int \frac{(d+e\,x^n)^q}{b-r+2\,c\,x^n}\,dx-\frac{2\,c}{r}\int \frac{(d+e\,x^n)^q}{b+r+2\,c\,x^n}\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_/(a_+b_.*x_^n_+c_.*x_^n2_),x_Symbol] :=
  With[{r=Rt[b^2-4*a*c,2]},
  2*c/r*Int[(d+e*x^n)^q/(b-r+2*c*x^n),x] - 2*c/r*Int[(d+e*x^n)^q/(b+r+2*c*x^n),x]] /;
FreeQ[{a,b,c,d,e,n,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[q]]
```

```
Int[(d_+e_.*x_^n_)^q_/(a_+c_.*x_^n2_),x_Symbol] :=
  With[{r=Rt[-a*c,2]},
  -c/(2*r)*Int[(d+e*x^n)^q/(r-c*x^n),x] - c/(2*r)*Int[(d+e*x^n)^q/(r+c*x^n),x]] /;
FreeQ[{a,c,d,e,n,q},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[q]]
```

9. $\int (d + e x^n) (a + b x^n + c x^{2n})^p \, dx$ when $b^2 - 4 a c \neq 0$

1: $\int (d + e x^n) (a + b x^n + c x^{2n})^p \, dx$ when $b^2 - 4 a c \neq 0 \land p < -1$

Derivation: Trinomial recurrence 2b with $m = 0$

Rule 1.2.3.3.9.1: If $b^2 - 4 a c \neq 0 \land p < -1$, then

$$\int (d + e x^n) (a + b x^n + c x^{2n})^p \, dx \longrightarrow$$

$$-\frac{x \left(d b^2 - a b e - 2 a c d + (b d - 2 a e) c x^n\right) (a + b x^n + c x^{2n})^{p+1}}{a n (p+1) (b^2 - 4 a c)} + \frac{1}{a n (p+1) (b^2 - 4 a c)} \cdot$$

$$\int \left((n p + n + 1) d b^2 - a b e - 2 a c d (2 n p + 2 n + 1) + (2 n p + 3 n + 1) (d b - 2 a e) c x^n\right) (a + b x^n + c x^{2n})^{p+1} \, dx$$

Program code:

```
Int[(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  -x*(d*b^2-a*b*e-2*a*c*d+(b*d-2*a*e)*c*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*n*(p+1)*(b^2-4*a*c)) +
  1/(a*n*(p+1)*(b^2-4*a*c))*
    Int[Simp[(n*p+n+1)*d*b^2-a*b*e-2*a*c*d*(2*n*p+2*n+1)+(2*n*p+3*n+1)*(d*b-2*a*e)*c*x^n,x]*
      (a+b*x^n+c*x^(2*n))^(p+1),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[p,-1]
```

```
Int[(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  -x*(d+e*x^n)*(a+c*x^(2*n))^(p+1)/(2*a*n*(p+1)) +
  1/(2*a*n*(p+1))*Int[(d*(2*n*p+2*n+1)+e*(2*n*p+3*n+1)*x^n)*(a+c*x^(2*n))^(p+1),x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n] && ILtQ[p,-1]
```

2: $\int (d+e\,x^n)\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Algebraic expansion

Rule 1.2.3.3.9.2: If $b^2-4\,a\,c\neq 0$, then

$$\int (d+e\,x^n)\,(a+b\,x^n+c\,x^{2\,n})^p\,dx \to \int \text{ExpandIntegrand}\left[(d+e\,x^n)\,(a+b\,x^n+c\,x^{2\,n})^p,\,x\right]dx$$

Program code:

```
Int[(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

```
Int[(d_+e_.*x_^n_)*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)*(a+c*x^(2*n))^p,x],x] /;
FreeQ[{a,c,d,e,n},x] && EqQ[n2,2*n]
```

10: $\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,p\in\mathbb{Z}^+\land\,2\,n\,p+n\,q+1\neq 0$

Reference: G&R 2.110.5, CRC 88a

Derivation: Binomial recurrence 3a

Note: This rule reduces the degree of the polynomial in the resulting integrand.

Rule 1.2.3.3.10: If $b^2-4\,a\,c\neq 0\,\land\,p\in\mathbb{Z}^+\land\,2\,n\,p+n\,q+1\neq 0$, then

$$\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx \to \int (d+e\,x^n)^q\,\left((a+b\,x^n+c\,x^{2\,n})^p-c^p\,x^{2\,n\,p}\right)dx+c^p\int x^{2\,n\,p}\,(d+e\,x^n)^q\,dx$$

$$\to \frac{c^p x^{2np-n+1} (d+e x^n)^{q+1}}{e (2np+nq+1)} + \int (d+e x^n)^q \left((a+b x^n+c x^{2n})^p - c^p x^{2np} - \frac{d c^p (2np-n+1) x^{2np-n}}{e (2np+nq+1)}\right) dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  c^p*x^(2*n*p-n+1)*(d+e*x^n)^(q+1)/(e*(2*n*p+n*q+1)) +
  Int[(d+e*x^n)^q*ExpandToSum[(a+b*x^n+c*x^(2*n))^p-c^p*x^(2*n*p)-d*c^p*(2*n*p-n+1)*x^(2*n*p-n)/(e*(2*n*p+n*q+1)),x],x] /;
FreeQ[{a,b,c,d,e,n,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && NeQ[2*n*p+n*q+1,0] && IGtQ[n,0] && Not[IGtQ[q,0]]
```

```
Int[(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  c^p*x^(2*n*p-n+1)*(d+e*x^n)^(q+1)/(e*(2*n*p+n*q+1)) +
  Int[(d+e*x^n)^q*ExpandToSum[(a+c*x^(2*n))^p-c^p*x^(2*n*p)-d*c^p*(2*n*p-n+1)*x^(2*n*p-n)/(e*(2*n*p+n*q+1)),x],x] /;
FreeQ[{a,c,d,e,n,q},x] && EqQ[n2,2*n] && IGtQ[p,0] && NeQ[2*n*p+n*q+1,0] && IGtQ[n,0] && Not[IGtQ[q,0]]
```

11: $\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land ((p \mid q) \in \mathbb{Z} \lor p \in \mathbb{Z}^+ \lor q \in \mathbb{Z}^+)$

Derivation: Algebraic expansion

Rule 1.2.3.3.11: If $b^2-4ac \neq 0 \land ((p \mid q) \in \mathbb{Z} \lor p \in \mathbb{Z}^+ \lor q \in \mathbb{Z}^+)$, then

$$\int (d+e x^n)^q (a+b x^n+c x^{2n})^p \, dx \to \int \text{ExpandIntegrand}\left[(d+e x^n)^q (a+b x^n+c x^{2n})^p, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  (IntegersQ[p,q] && Not[IntegerQ[n]] || IGtQ[p,0] || IGtQ[q,0] && Not[IntegerQ[n]])
```

```
Int[(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^n)^q*(a+c*x^(2*n))^p,x],x] /;
FreeQ[{a,c,d,e,n,p,q},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] &&
  (IntegersQ[p,q] && Not[IntegerQ[n]] || IGtQ[p,0] || IGtQ[q,0] && Not[IntegerQ[n]])
```

12: $\int (d+e\,x^n)^q\,(a+c\,x^{2\,n})^p\,dx$ when $c\,d^2+a\,e^2\neq 0\,\land\,p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $q\in\mathbb{Z}$, then $(d+e\,x^n)^q==\left(\frac{d}{d^2-e^2\,x^{2\,n}}-\frac{e\,x^n}{d^2-e^2\,x^{2\,n}}\right)^{-q}$

Note: Resulting integrands are of the form $x^m\,(a+b\,x^{2\,n})^p\,(c+d\,x^{2\,n})^q$ which are integrable in terms of the Appell hypergeometric function .

Rule 1.2.3.3.12: If $c\,d^2+a\,e^2\neq 0\,\land\,p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-$, then

$$\int (d+e\,x^n)^q\,(a+c\,x^{2\,n})^p\,dx\,\longrightarrow\,\int (a+c\,x^{2\,n})^p\,\text{ExpandIntegrand}\left[\left(\frac{d}{d^2-e^2\,x^{2\,n}}-\frac{e\,x^n}{d^2-e^2\,x^{2\,n}}\right)^{-q},\,x\right]dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+c*x^(2*n))^p,(d/(d^2-e^2*x^(2*n))-e*x^n/(d^2-e^2*x^(2*n)))^(-q),x],x] /;
FreeQ[{a,c,d,e,n,p},x] && EqQ[n2,2*n] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[q,0]
```

U: $\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$

Rule 1.2.3.3.X:

$$\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx\;\longrightarrow\;\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  Unintegrable[(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n]
```

```
Int[(d_+e_.*x_^n_)^q_*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  Unintegrable[(d+e*x^n)^q*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,n,p,q},x] && EqQ[n2,2*n]
```

S: $\int (d+e\,u^n)^q\,(a+b\,u^n+c\,u^{2\,n})^p\,dx$ when $u == f+g\,x$

Derivation: Integration by substitution

Rule 1.2.3.3.S: If $u == f+g\,x$, then

$$\int (d+e\,u^n)^q\,(a+b\,u^n+c\,u^{2\,n})^p\,dx\;\longrightarrow\;\frac{1}{g}\,\text{Subst}\!\left[\int (d+e\,x^n)^q\,(a+b\,x^n+c\,x^{2\,n})^p\,dx,\;x,\;u\right]$$

Program code:

```
Int[(d_+e_.*u_^n_)^q_.*(a_+b_.*u_^n_+c_.*u_^n2_)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x],x,u] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(d_+e_.*u_^n_)^q_.*(a_+c_.*u_^n2_)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(d+e*x^n)^q*(a+c*x^(2*n))^p,x],x,u] /;
FreeQ[{a,c,d,e,n,p,q},x] && EqQ[n2,2*n] && LinearQ[u,x] && NeQ[u,x]
```

Rules for integrands of the form $(d + e\, x^{-n})^q \left(a + b\, x^n + c\, x^{2\,n}\right)^p$

1. $\int (d + e\, x^{-n})^q \left(a + b\, x^n + c\, x^{2\,n}\right)^p \, dx$ when $p \in \mathbb{Z} \lor q \in \mathbb{Z}$

1: $\int (d + e\, x^{-n})^q \left(a + b\, x^n + c\, x^{2\,n}\right)^p \, dx$ when $q \in \mathbb{Z} \land (n > 0 \lor p \notin \mathbb{Z})$

Derivation: Algebraic simplification

Basis: If $q \in \mathbb{Z}$, then $(d + e\, x^{-n})^q == x^{-n\,q}\,(e + d\, x^n)^q$

Rule: If $q \in \mathbb{Z} \land (n > 0 \lor p \notin \mathbb{Z})$, then

$$\int (d + e\, x^{-n})^q \left(a + b\, x^n + c\, x^{2\,n}\right)^p \, dx \longrightarrow \int x^{-n\,q}\,(e + d\, x^n)^q \left(a + b\, x^n + c\, x^{2\,n}\right)^p \, dx$$

Program code:

```
Int[(d_+e_.*x_^mn_.)^q_.*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(-n*q)*(e+d*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[n2,2*n] && EqQ[mn,-n] && IntegerQ[q] && (PosQ[n] || Not[IntegerQ[p]])
```

```
Int[(d_+e_.*x_^mn_.)^q_.*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[x^(mn*q)*(e+d*x^(-mn))^q*(a+c*x^n2)^p,x] /;
FreeQ[{a,c,d,e,mn,p},x] && EqQ[n2,-2*mn] && IntegerQ[q] && (PosQ[n2] || Not[IntegerQ[p]])
```

2: $\int (d+e\,x^n)^q\,\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p\in\mathbb{Z}$, then $\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p == x^{-2\,n\,p}\,\left(c+b\,x^n+a\,x^{2\,n}\right)^p$

Rule: If $p\in\mathbb{Z}$, then

$$\int (d+e\,x^n)^q\,\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p\,dx \,\longrightarrow\, \int x^{-2\,n\,p}\,(d+e\,x^n)^q\,\left(c+b\,x^n+a\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_.)^q_.*(a_.+b_.*x_^mn_.+c_.*x_^mn2_.)^p_.,x_Symbol] :=
  Int[x^(-2*n*p)*(d+e*x^n)^q*(c+b*x^n+a*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n,q},x] && EqQ[mn,-n] && EqQ[mn2,2*mn] && IntegerQ[p]
```

```
Int[(d_+e_.*x_^n_.)^q_.*(a_.+c_.*x_^mn2_.)^p_.,x_Symbol] :=
  Int[x^(-2*n*p)*(d+e*x^n)^q*(c+a*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,n,q},x] && EqQ[mn2,-2*n] && IntegerQ[p]
```

2. $\int \left(d+e\,x^{-n}\right)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,q\notin\mathbb{Z}$

1: $\int \left(d+e\,x^{-n}\right)^q\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,q\notin\mathbb{Z}\,\land\,n>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,q}\,\left(d+e\,x^{-n}\right)^q}{\left(1+\frac{d\,x^n}{e}\right)^q} == 0$

Rule: If $p\notin\mathbb{Z}\,\land\,q\notin\mathbb{Z}\,\land\,n>0$, then

$$\int \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \;\longrightarrow\; \frac{e^{\text{IntPart}[q]}\,x^{n\,\text{FracPart}[q]}\left(d+e\,x^{-n}\right)^{\text{FracPart}[q]}}{\left(1+\frac{d\,x^n}{e}\right)^{\text{FracPart}[q]}}\int x^{-n\,q}\left(1+\frac{d\,x^n}{e}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^mn_.)^q_*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  e^IntPart[q]*x^(n*FracPart[q])*(d+e*x^(-n))^FracPart[q]/(1+d*x^n/e)^FracPart[q]*Int[x^(-n*q)*(1+d*x^n/e)^q*(a+b*x^n+c*x^(2*n))^p,x
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && EqQ[mn,-n] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n]
```

```
Int[(d_+e_.*x_^mn_.)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  e^IntPart[q]*x^(-mn*FracPart[q])*(d+e*x^mn)^FracPart[q]/(1+d*x^(-mn)/e)^FracPart[q]*Int[x^(mn*q)*(1+d*x^(-mn)/e)^q*(a+c*x^n2)^p,x
FreeQ[{a,c,d,e,mn,p,q},x] && EqQ[n2,-2*mn] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n2]
```

x: $\int \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $p\notin\mathbb{Z}\land q\notin\mathbb{Z}\land n>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,q}\,(d+e\,x^{-n})^q}{(e+d\,x^n)^q} == 0$

Rule: If $p\notin\mathbb{Z}\land q\notin\mathbb{Z}\land n>0$, then

$$\int \left(d+e\,x^{-n}\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \;\longrightarrow\; \frac{x^{n\,\text{FracPart}[q]}\left(d+e\,x^{-n}\right)^{\text{FracPart}[q]}}{\left(e+d\,x^n\right)^{\text{FracPart}[q]}}\int x^{-n\,q}\left(e+d\,x^n\right)^q \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$$

Program code:

```
(* Int[(d_+e_.*x_^mn_.)^q_*(a_.+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  x^(n*FracPart[q])*(d+e*x^(-n))^FracPart[q]/(e+d*x^n)^FracPart[q]*Int[x^(-n*q)*(e+d*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[n2,2*n] && EqQ[mn,-n] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n] *)
```

```
(* Int[(d_+e_.*x_^mn_.)^q_*(a_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  x^(-mn*FracPart[q])*(d+e*x^mn)^FracPart[q]/(e+d*x^(-mn))^FracPart[q]*Int[x^(mn*q)*(e+d*x^(-mn))^q*(a+c*x^n2)^p,x] /;
FreeQ[{a,c,d,e,mn,p,q},x] && EqQ[n2,-2*mn] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n2] *)
```

2: $\int (d + e\,x^n)^q\,(a + b\,x^{-n} + c\,x^{-2\,n})^p\,dx$ when $p \notin \mathbb{Z} \land q \notin \mathbb{Z} \land n > 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{2\,n\,p}\,(a+b\,x^{-n}+c\,x^{-2\,n})^p}{(c+b\,x^n+a\,x^{2\,n})^p} == 0$

Rule: If $p \notin \mathbb{Z} \land q \notin \mathbb{Z} \land n > 0$, then

$$\int (d + e\,x^n)^q\,(a + b\,x^{-n} + c\,x^{-2\,n})^p\,dx \to \frac{x^{2\,n\,\text{FracPart}[p]}\,(a + b\,x^{-n} + c\,x^{-2\,n})^{\text{FracPart}[p]}}{(c + b\,x^n + a\,x^{2\,n})^{\text{FracPart}[p]}} \int x^{-2\,n\,p}\,(d + e\,x^n)^q\,(c + b\,x^n + a\,x^{2\,n})^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_.)^q_.*(a_.+b_.*x_^mn_.+c_.*x_^mn2_.)^p_,x_Symbol] :=
  x^(2*n*FracPart[p])*(a+b*x^(-n)+c*x^(-2*n))^FracPart[p]/(c+b*x^n+a*x^(2*n))^FracPart[p]*
    Int[x^(-2*n*p)*(d+e*x^n)^q*(c+b*x^n+a*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[mn,-n] && EqQ[mn2,2*mn] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n]
```

```
Int[(d_+e_.*x_^n_.)^q_.*(a_.+c_.*x_^mn2_.)^p_,x_Symbol] :=
  x^(2*n*FracPart[p])*(a+c*x^(-2*n))^FracPart[p]/(c+a*x^(2*n))^FracPart[p]*
    Int[x^(-2*n*p)*(d+e*x^n)^q*(c+a*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,n,p,q},x] && EqQ[mn2,-2*n] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && PosQ[n]
```

Rules for integrands of the form $(d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p$

1: $\int (d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $a+b\,x^{-n}+c\,x^n == x^{-n}\,(b+a\,x^n+c\,x^{2\,n})$

Rule: If $p\in\mathbb{Z}$, then

$$\int (d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx \longrightarrow \int x^{-n\,p}\,(d+e\,x^n)^q\,(b+a\,x^n+c\,x^{2\,n})^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  Int[x^(-n*p)*(d+e*x^n)^q*(b+a*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n,q},x] && EqQ[mn,-n] && IntegerQ[p]
```

2: $\int (d+e\,x^n)^q\,(a+b\,x^{-n}+c\,x^n)^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,p}\,(a+b\,x^{-n}+c\,x^n)^p}{(b+a\,x^n+c\,x^{2\,n})^p} == 0$

Basis: $\frac{x^{n\,p}\,(a+b\,x^{-n}+c\,x^n)^p}{(b+a\,x^n+c\,x^{2\,n})^p} == \frac{x^{n\,\mathrm{FracPart}[p]}\,(a+b\,x^{-n}+c\,x^n)^{\mathrm{FracPart}[p]}}{(b+a\,x^n+c\,x^{2\,n})^{\mathrm{FracPart}[p]}}$

Rule: If $p\notin\mathbb{Z}$, then

$$\int (d+e x^n)^q \left(a+b x^{-n}+c x^n\right)^p dx \longrightarrow \frac{x^{n\,\text{FracPart}[p]} \left(a+b x^{-n}+c x^n\right)^{\text{FracPart}[p]}}{\left(b+a x^n+c x^{2n}\right)^{\text{FracPart}[p]}} \int x^{-n p} (d+e x^n)^q \left(b+a x^n+c x^{2n}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  x^(n*FracPart[p])*(a+b/x^n+c*x^n)^FracPart[p]/(b+a*x^n+c*x^(2*n))^FracPart[p]*
    Int[x^(-n*p)*(d+e*x^n)^q*(b+a*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,n,p,q},x] && EqQ[mn,-n] && Not[IntegerQ[p]]
```

Rules for integrands of the form $(d+e x^n)^q (f+g x^n)^r \left(a+b x^n+c x^{2n}\right)^p$

1: $\int (d+e x^n)^q (f+g x^n)^r \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4 a c == 0 \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4 a c == 0$, then $\partial_x \frac{\left(a+b x^n+c x^{2n}\right)^p}{(b+2 c x^n)^{2p}} == 0$

Basis: If $b^2-4 a c == 0$, then $\frac{\left(a+b x^n+c x^{2n}\right)^p}{(b+2 c x^n)^{2p}} == \frac{\left(a+b x^n+c x^{2n}\right)^{\text{FracPart}[p]}}{(4 c)^{\text{IntPart}[p]} (b+2 c x^n)^{2\,\text{FracPart}[p]}}$

Rule: If $b^2-4 a c == 0 \wedge 2 p \notin \mathbb{Z}$, then

$$\int (d+e x^n)^q (f+g x^n)^r \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \frac{\left(a+b x^n+c x^{2n}\right)^{\text{FracPart}[p]}}{(4 c)^{\text{IntPart}[p]} (b+2 c x^n)^{2\,\text{FracPart}[p]}} \int (d+e x^n)^q (f+g x^n)^r (b+2 c x^n)^{2p} dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(f_+g_.*x_^n_)^r_.*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x^n)^(2*FracPart[p]))*
    Int[(d+e*x^n)^q*(f+g*x^n)^r*(b+2*c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q,r},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int (d+e x^n)^q (f+g x^n)^r (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac\neq 0 \wedge c d^2-b d e+a e^2==0$

1: $\int (d+e x^n)^q (f+g x^n)^r (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac\neq 0 \wedge c d^2-b d e+a e^2==0 \wedge p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c d^2-b d e+a e^2==0$, then $a+b z+c z^2==(d+e z)\left(\frac{a}{d}+\frac{c z}{e}\right)$

Rule: If $b^2-4ac\neq 0 \wedge c d^2-b d e+a e^2==0 \wedge p\in\mathbb{Z}$, then

$$\int (d+e x^n)^q (f+g x^n)^r (a+b x^n+c x^{2n})^p \, dx \longrightarrow \int (d+e x^n)^{p+q} (f+g x^n)^r \left(\frac{a}{d}+\frac{c x^n}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(f_+g_.*x_^n_)^r_.*(a_+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(d+e*x^n)^(p+q)*(f+g*x^n)^r*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,n,q,r},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_^n_)^q_.*(f_+g_.*x_^n_)^r_.*(a_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(d+e*x^n)^(p+q)*(f+g*x^n)^r*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,c,d,e,f,g,n,q,r},x] && EqQ[n2,2*n] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

2: $\int (d+e x^n)^q (f+g x^n)^r (a+b x^n+c x^{2n})^p \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c d^2-b d e+a e^2 == 0$, then $\partial_x \frac{(a+b x^n+c x^{2n})^p}{(d+e x^n)^p \left(\frac{a}{d}+\frac{c x^n}{e}\right)^p} == 0$

Basis: If $c d^2-b d e+a e^2 == 0$, then $\frac{(a+b x^n+c x^{2n})^p}{(d+e x^n)^p \left(\frac{a}{d}+\frac{c x^n}{e}\right)^p} == \frac{(a+b x^n+c x^{2n})^{\text{FracPart}[p]}}{(d+e x^n)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^n}{e}\right)^{\text{FracPart}[p]}}$

Rule: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z}$, then

$$\int (d+e x^n)^q (f+g x^n)^r (a+b x^n+c x^{2n})^p \, dx \to \frac{(a+b x^n+c x^{2n})^{\text{FracPart}[p]}}{(d+e x^n)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^n}{e}\right)^{\text{FracPart}[p]}} \int (d+e x^n)^{p+q} (f+g x^n)^r \left(\frac{a}{d}+\frac{c x^n}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^n_)^q_.*(f_+g_.*x_^n_)^r_.*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/((d+e*x^n)^FracPart[p]*(a/d+(c*x^n)/e)^FracPart[p])*
    Int[(d+e*x^n)^(p+q)*(f+g*x^n)^r*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q,r},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]]
```

```
Int[(d_+e_.*x_^n_)^q_.*(f_+g_.*x_^n_)^r_.*(a_+c_.*x_^n2_)^p_,x_Symbol] :=
  (a+c*x^(2*n))^FracPart[p]/((d+e*x^n)^FracPart[p]*(a/d+(c*x^n)/e)^FracPart[p])*
    Int[(d+e*x^n)^(p+q)*(f+g*x^n)^r*(a/d+c/e*x^n)^p,x] /;
FreeQ[{a,c,d,e,f,g,n,p,q,r},x] && EqQ[n2,2*n] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]]
```

3. $\int (d_1 + e_1 x^{n/2})^q (d_2 + e_2 x^{n/2})^q (a + b x^n + c x^{2n})^p \, dx$ when $d_2 e_1 + d_1 e_2 == 0$

1: $\int (d_1 + e_1 x^{n/2})^q (d_2 + e_2 x^{n/2})^q (a + b x^n + c x^{2n})^p \, dx$ when $d_2 e_1 + d_1 e_2 == 0 \land (q \in \mathbb{Z} \lor d_1 > 0 \land d_2 > 0)$

Derivation: Algebraic simplification

Basis: If $d_2 e_1 + d_1 e_2 == 0 \land (q \in \mathbb{Z} \lor d_1 > 0 \land d_2 > 0)$, then $(d_1 + e_1 x^{n/2})^q (d_2 + e_2 x^{n/2})^q == (d_1 d_2 + e_1 e_2 x^n)^q$

Rule: If $d_2 e_1 + d_1 e_2 == 0 \land (q \in \mathbb{Z} \lor d_1 > 0 \land d_2 > 0)$, then

$$\int (d_1 + e_1 x^{n/2})^q (d_2 + e_2 x^{n/2})^q (a + b x^n + c x^{2n})^p \, dx \to \int (d_1 d_2 + e_1 e_2 x^n)^q (a + b x^n + c x^{2n})^p \, dx$$

Program code:

```
Int[(d1_+e1_.*x_^non2_.)^q_.*(d2_+e2_.*x_^non2_.)^q_.*(a_.+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  Int[(d1*d2+e1*e2*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n,p,q},x] && EqQ[n2,2*n] && EqQ[non2,n/2] && EqQ[d2*e1+d1*e2,0] && (IntegerQ[q] || GtQ[d1,0] && GtQ[d2,0])
```

2: $\int (d_1 + e_1 x^{n/2})^q (d_2 + e_2 x^{n/2})^q (a + b x^n + c x^{2n})^p \, dx$ when $d_2 e_1 + d_1 e_2 == 0$

Derivation: Piecewise constant extraction

Basis: If $d_2 e_1 + d_1 e_2 == 0$, then $\partial_x \frac{(d_1+e_1 x^{n/2})^q (d_2+e_2 x^{n/2})^q}{(d_1 d_2+e_1 e_2 x^n)^q} == 0$

Rule: If $d_2 e_1 + d_1 e_2 == 0$, then

$$\int (d_1 + e_1 x^{n/2})^q (d_2 + e_2 x^{n/2})^q (a + b x^n + c x^{2n})^p \, dx \to \frac{(d_1 + e_1 x^{n/2})^{\text{FracPart}[q]} (d_2 + e_2 x^{n/2})^{\text{FracPart}[q]}}{(d_1 d_2 + e_1 e_2 x^n)^{\text{FracPart}[q]}} \int (d_1 d_2 + e_1 e_2 x^n)^q (a + b x^n + c x^{2n})^p \, dx$$

Program code:

```
Int[(d1_+e1_.*x_^non2_.)^q_.*(d2_+e2_.*x_^non2_.)^q_.*(a_.+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  (d1+e1*x^(n/2))^FracPart[q]*(d2+e2*x^(n/2))^FracPart[q]/(d1*d2+e1*e2*x^n)^FracPart[q]*
    Int[(d1*d2+e1*e2*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n,p,q},x] && EqQ[n2,2*n] && EqQ[non2,n/2] && EqQ[d2*e1+d1*e2,0]
```

Rules for integrands of the form $(A + B x^m) (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p$

1: $\int (A + B x^m) (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p \, dx$ when $m - n + 1 == 0$

Derivation: Algebraic expansion

Rule: If $m - n + 1 == 0$, then

$$\int (A + B x^m) (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p \, dx \longrightarrow A \int (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p \, dx + B \int x^m (d + e x^n)^q \left(a + b x^n + c x^{2n}\right)^p \, dx$$

Program code:

```
Int[(A_+B_.*x_^m_.)*(d_+e_.*x_^n_)^q_.*(a_+b_.*x_^n_+c_.*x_^n2_)^p_.,x_Symbol] :=
  A*Int[(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] + B*Int[x^m*(d+e*x^n)^q*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,e,A,B,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[m-n+1,0]
```

```
Int[(A_+B_.*x_^m_.)*(d_+e_.*x_^n_)^q_.*(a_+c_.*x_^n2_)^p_.,x_Symbol] :=
  A*Int[(d+e*x^n)^q*(a+c*x^(2*n))^p,x] + B*Int[x^m*(d+e*x^n)^q*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,c,d,e,A,B,m,n,p,q},x] && EqQ[n2,2*n] && EqQ[m-n+1,0]
```