Rules for integrands of the form $(d + e\,x^{n-q})\,(a\,x^q + b\,x^n + c\,x^{2\,n-q})^p$

1: $\int (A + B\,x^{n-q})\,(a\,x^q + b\,x^n + c\,x^{2\,n-q})^p\,dx$ when $p \in \mathbb{Z} \land q < n$

Rule: If $p \in \mathbb{Z} \land q < n$, then

$$\int (A + B\,x^{n-q})\,(a\,x^q + b\,x^n + c\,x^{2\,n-q})^p\,dx \longrightarrow \int x^{p\,q}\,(A + B\,x^{n-q})\,(a + b\,x^{n-q} + c\,x^{2\,(n-q)})^p\,dx$$

Program code:

```
Int[(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  Int[x^(p*q)*(A+B*x^(n-q))*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,A,B,n,q},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && IntegerQ[p] && PosQ[n-q]
```

x. $\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx$ when $q < n \land p + \frac{1}{2} \in \mathbb{Z}$

x: $\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx$ when $q < n \land p + \frac{1}{2} \in \mathbb{Z}^+$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a x^q + b x^n + c x^{2n-q}}}{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2(n-q)}}} == 0$

Rule: If $q < n \land p + \frac{1}{2} \in \mathbb{Z}^+$, then

$$\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx \to \frac{\sqrt{a x^q + b x^n + c x^{2n-q}}}{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2(n-q)}}} \int x^{q p} (A + B x^{n-q}) (a + b x^{n-q} + c x^{2(n-q)})^p \, dx$$

Program code:

```
(* Int[(A_+B_.*x_^j_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]/(x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))])*
    Int[x^(q*p)*(A+B*x^(n-q))*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,A,B,n,p,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && PosQ[n-q] && IGtQ[p+1/2,0] *)
```

x: $\int \left(A+B x^{n-q}\right) \left(a x^q+b x^n+c x^{2 n-q}\right)^p dx$ when $q<n \land p-\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{q/2}\sqrt{a+b x^{n-q}+c x^{2 (n-q)}}}{\sqrt{a x^q+b x^n+c x^{2 n-q}}} == 0$

Rule: If $q<n \land p-\frac{1}{2}\in\mathbb{Z}^-$, then

$$\int \left(A+B x^{n-q}\right) \left(a x^q+b x^n+c x^{2 n-q}\right)^p dx \longrightarrow \frac{x^{q/2}\sqrt{a+b x^{n-q}+c x^{2 (n-q)}}}{\sqrt{a x^q+b x^n+c x^{2 n-q}}}\int x^{q p} \left(A+B x^{n-q}\right) \left(a+b x^{n-q}+c x^{2 (n-q)}\right)^p dx$$

Program code:

```
(* Int[(A_+B_.*x_^j_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]/Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]*
    Int[x^(q*p)*(A+B*x^(n-q))*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,A,B,n,p,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && PosQ[n-q] && ILtQ[p-1/2,0] *)
```

x: $\int \left(A + B x^{n-q}\right) \sqrt{a x^q + b x^n + c x^{2 n-q}} \, dx$ when $q < n$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a x^q+b x^n+c x^{2 n-q}}}{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2 (n-q)}}} == 0$

Rule: If $q < n$, then

$$\int \left(A + B x^{n-q}\right) \sqrt{a x^q + b x^n + c x^{2 n-q}} \, dx \to \frac{\sqrt{a x^q + b x^n + c x^{2 n-q}}}{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2 (n-q)}}} \int x^{q/2} \left(A + B x^{n-q}\right) \sqrt{a + b x^{n-q} + c x^{2 (n-q)}} \, dx$$

Program code:

```
(* Int[(A_+B_.*x_^j_.)*Sqrt[a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]/(x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))])*
    Int[x^(q/2)*(A+B*x^(n-q))*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))],x] /;
FreeQ[{a,b,c,A,B,n,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && PosQ[n-q] *)
```

2: $\int \frac{A + B x^{n-q}}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \, dx$ when $q < n$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2(n-q)}}}{\sqrt{a x^q+b x^n+c x^{2n-q}}} == 0$

Rule: If $q < n$, then

$$\int \frac{A + B x^{n-q}}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \, dx \longrightarrow \frac{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2(n-q)}}}{\sqrt{a x^q + b x^n + c x^{2n-q}}} \int \frac{A + B x^{n-q}}{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2(n-q)}}} \, dx$$

Program code:

```
Int[(A_+B_.*x_^j_.)/Sqrt[a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]/Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]*
    Int[(A+B*x^(n-q))/(x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]),x] /;
FreeQ[{a,b,c,A,B,n,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && PosQ[n-q] && EqQ[n,3] && EqQ[q,2]
```

3: $\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx$ when $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land p > 0 \land p (2n - q) + 1 \neq 0 \land p q + (n - q)(2p + 1) + 1 \neq 0$

Derivation: Trinomial recurrence 1b with $m = 0$

Rule: If $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land p > 0 \land p (2n - q) + 1 \neq 0 \land p q + (n - q)(2p + 1) + 1 \neq 0$, then

$$\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx \longrightarrow$$

$$\left(\left(x \left(b B (n - q) p + A c (p q + (n - q)(2p + 1) + 1) + B c (p (2n - q) + 1) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p\right) / \left(c (p (2n - q) + 1) (p q + (n - q)(2p + 1) + 1)\right)\right) + \frac{(n - q) p}{c (p (2n - q) + 1) (p q + (n - q)(2p + 1) + 1)} \cdot$$

$$\int x^q \left(2 a A c (p q + (n - q)(2p + 1) + 1) - a b B (p q + 1) + \left(2 a B c (p (2n - q) + 1) + A b c (p q + (n - q)(2p + 1) + 1) - b^2 B (p q + (n - q) p + 1)\right) x^{n-q}\right) \cdot$$

$$\left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^{p-1}\,dx$$

Program code:

```
Int[(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_,x_Symbol] :=
  x*(b*B*(n-q)*p+A*c*(p*q+(n-q)*(2*p+1)+1)+B*c*(p*(2*n-q)+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p/
    (c*(p*(2*n-q)+1)*(p*q+(n-q)*(2*p+1)+1)) +
  (n-q)*p/(c*(p*(2*n-q)+1)*(p*q+(n-q)*(2*p+1)+1))*
    Int[x^q*
      (2*a*A*c*(p*q+(n-q)*(2*p+1)+1)-a*b*B*(p*q+1)+(2*a*B*c*(p*(2*n-q)+1)+A*b*c*(p*q+(n-q)*(2*p+1)+1)-b^2*B*(p*q+(n-q)*p+1))*x^(n-q))*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c,A,B,n,q},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && GtQ[p,0] &&
  NeQ[p*(2*n-q)+1,0] && NeQ[p*q+(n-q)*(2*p+1)+1,0]
```

```
Int[(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_,x_Symbol] :=
  With[{n=q+r},
  x*(A*(p*q+(n-q)*(2*p+1)+1)+B*(p*(2*n-q)+1)*x^(n-q))*(a*x^q+c*x^(2*n-q))^p/((p*(2*n-q)+1)*(p*q+(n-q)*(2*p+1)+1)) +
  (n-q)*p/((p*(2*n-q)+1)*(p*q+(n-q)*(2*p+1)+1))*
    Int[x^q*(2*a*A*(p*q+(n-q)*(2*p+1)+1)+(2*a*B*(p*(2*n-q)+1))*x^(n-q))*(a*x^q+c*x^(2*n-q))^(p-1),x] /;
 EqQ[j,2*n-q] && NeQ[p*(2*n-q)+1,0] && NeQ[p*q+(n-q)*(2*p+1)+1,0]] /;
FreeQ[{a,c,A,B,q},x] && Not[IntegerQ[p]] && GtQ[p,0]
```

4: $\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx$ when $p \notin \mathbb{Z} \wedge b^2 - 4ac \neq 0 \wedge p < -1$

Derivation: Trinomial recurrence 2b with $m = 0$

Rule: If $p \notin \mathbb{Z} \wedge b^2 - 4ac \neq 0 \wedge p < -1$, then

$$\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx \to$$

$$-\left(\left(x^{-q+1} \left(A b^2 - a b B - 2 a A c + (A b - 2 a B) c x^{n-q}\right) (a x^q + b x^n + c x^{2n-q})^{p+1}\right) / \left(a (n-q) (p+1) (b^2 - 4ac)\right)\right) + \frac{1}{a (n-q) (p+1) (b^2 - 4ac)} \cdot$$

$$\int x^{-q} \left(A b^2 (p q + (n-q)(p+1) + 1) - a b B (p q + 1) - 2 a A c (p q + 2 (n-q)(p+1) + 1) + (p q + (n-q)(2p+3) + 1) (A b - 2 a B) c x^{n-q}\right) (a x^q + b x^n + c x^{2n-q})^{p+1} \, dx$$

Program code:

```
Int[(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_,x_Symbol] :=
  -x^(-q+1)*(A*b^2-a*b*B-2*a*A*c+(A*b-2*a*B)*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(a*(n-q)*(p+1)*(b^2-4*a*c)) +
  1/(a*(n-q)*(p+1)*(b^2-4*a*c))*
    Int[x^(-q)*
      ((A*b^2*(p*q+(n-q)*(p+1)+1)-a*b*B*(p*q+1)-2*a*A*c*(p*q+2*(n-q)*(p+1)+1)+(p*q+(n-q)*(2*p+3)+1)*(A*b-2*a*B)*c*x^(n-q))*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p+1)),x] /;
FreeQ[{a,b,c,A,B,n,q},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && LtQ[p,-1]
```

```
Int[(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_,x_Symbol] :=
  With[{n=q+r},
  -x^(-q+1)*(a*A*c+a*B*c*x^(n-q))*(a*x^q+c*x^(2*n-q))^(p+1)/(a*(n-q)*(p+1)*(2*a*c)) +
  1/(a*(n-q)*(p+1)*(2*a*c))*
    Int[x^(-q)*((a*A*c*(p*q+2*(n-q)*(p+1)+1)+a*B*c*(p*q+(n-q)*(2*p+3)+1)*x^(n-q))*(a*x^q+c*x^(2*n-q))^(p+1)),x] /;
 EqQ[j,2*n-q]] /;
FreeQ[{a,c,A,B,q},x] && Not[IntegerQ[p]] && LtQ[p,-1]
```

x: $\int (A + B x^{k-j}) (a x^j + b x^k + c x^{2k-j})^p \, dx$ when $k > j \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a x^j + b x^k + c x^{2k-j})^p}{x^{jp} (a + b x^{k-j} + c x^{2(k-j)})^p} == 0$

Rule: If $k > j \wedge p \notin \mathbb{Z}$, then

$$\int x^m (A + B x^{k-j}) (a x^j + b x^k + c x^{2k-j})^p \, dx \longrightarrow \frac{(a x^j + b x^k + c x^{2k-j})^p}{x^{jp} (a + b x^{k-j} + c x^{2(k-j)})^p} \int x^{m+jp} (A + B x^{k-j}) (a + b x^{k-j} + c x^{2(k-j)})^p \, dx$$

Program code:

```
(* Int[(A_+B_.*x_^q_)*(a_.*x_^j_.+b_.*x_^k_.+c_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^k+c*x^n)^p/(x^(j*p)*(a+b*x^(k-j)+c*x^(2*(k-j)))^p)*
    Int[x^(j*p)*(A+B*x^(k-j))*(a+b*x^(k-j)+c*x^(2*(k-j)))^p,x] /;
FreeQ[{a,b,c,A,B,j,k,p},x] && EqQ[q,k-j] && EqQ[n,2*k-j] && PosQ[k-j] && Not[IntegerQ[p]] *)
```

X: $\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx$

Rule:

$$\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx \longrightarrow \int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx$$

Program code:

```
Int[(A_+B_.*x_^j_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_.,x_Symbol] :=
  Unintegrable[(A+B*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c,A,B,n,p,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q]
```

S: $\int (A + B u^{n-q}) (a u^q + b u^n + c u^{2n-q})^p \, dx$ when $u == d + e x$

Derivation: Integration by substitution

Rule: If $u == d + e x$, then

$$\int (A + B u^{n-q}) (a u^q + b u^n + c u^{2n-q})^p \, dx \to \frac{1}{e} \text{Subst}\left[\int (A + B x^{n-q}) (a x^q + b x^n + c x^{2n-q})^p \, dx, x, u\right]$$

Program code:

```
Int[(A_+B_.*u_^j_.)*(a_.*u_^q_.+b_.*u_^n_.+c_.*u_^r_.)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(A+B*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p,x],x,u] /;
FreeQ[{a,b,c,A,B,n,p,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && LinearQ[u,x] && NeQ[u,x]
```