Rules for integrands of the form $(a + b\,\mathrm{Sin}[c + d\,(e + f\,x)^n])^p$

1. $\int (a + b\,\mathrm{Sin}[c + d\,(e + f\,x)^n])^p\,dx$ when $p \in \mathbb{Z}^+ \land n \in \mathbb{Z}$

   1. $\int (a + b\,\mathrm{Sin}[c + d\,(e + f\,x)^n])^p\,dx$ when $p \in \mathbb{Z}^+ \land n - 1 \in \mathbb{Z}^+$

      1. $\int \mathrm{Sin}[c + d\,(e + f\,x)^n]\,dx$ when $n - 1 \in \mathbb{Z}^+$

         1. $\int \mathrm{Sin}[c + d\,(e + f\,x)^2]\,dx$

            1: $\int \mathrm{Sin}[d\,(e + f\,x)^2]\,dx$

Derivation: Primitive rule

Basis: $\mathrm{FresnelS}'[z] == \mathrm{Sin}\left[\frac{\pi z^2}{2}\right]$

Rule:

$$\int \mathrm{Sin}[d\,(e + f\,x)^2]\,dx \longrightarrow \frac{\sqrt{\frac{\pi}{2}}}{f\,\sqrt{d}}\,\mathrm{FresnelS}\left[\sqrt{\frac{2}{\pi}}\,\sqrt{d}\,(e + f\,x)\right]$$

Program code:

```
Int[Sin[d_.*(e_.+f_.*x_)^2],x_Symbol] :=
  Sqrt[Pi/2]/(f*Rt[d,2])*FresnelS[Sqrt[2/Pi]*Rt[d,2]*(e+f*x)] /;
FreeQ[{d,e,f},x]
```

```
Int[Cos[d_.*(e_.+f_.*x_)^2],x_Symbol] :=
  Sqrt[Pi/2]/(f*Rt[d,2])*FresnelC[Sqrt[2/Pi]*Rt[d,2]*(e+f*x)] /;
FreeQ[{d,e,f},x]
```

2: $\int \text{Sin}\left[c+d\,(e+f\,x)^2\right] dx$

Derivation: Algebraic expansion

Basis: $\text{Sin}[w+z] == \text{Sin}[w]\,\text{Cos}[z]+\text{Cos}[w]\,\text{Sin}[z]$

Basis: $\text{Cos}[w+z] == \text{Cos}[w]\,\text{Cos}[z]-\text{Sin}[w]\,\text{Sin}[z]$

Note: Although not essential, this rule produces antiderivatives in terms of Fresnel integrals instead of complex error functions.

Rule:

$$\int \text{Sin}\left[c+d\,(e+f\,x)^2\right] dx \rightarrow \text{Sin}[c]\int \text{Cos}\left[d\,(e+f\,x)^2\right] dx+\text{Cos}[c]\int \text{Sin}\left[d\,(e+f\,x)^2\right] dx$$

Program code:

```
Int[Sin[c_+d_.*(e_.+f_.*x_)^2],x_Symbol] :=
  Sin[c]*Int[Cos[d*(e+f*x)^2],x] + Cos[c]*Int[Sin[d*(e+f*x)^2],x] /;
FreeQ[{c,d,e,f},x]
```

```
Int[Cos[c_+d_.*(e_.+f_.*x_)^2],x_Symbol] :=
  Cos[c]*Int[Cos[d*(e+f*x)^2],x] - Sin[c]*Int[Sin[d*(e+f*x)^2],x] /;
FreeQ[{c,d,e,f},x]
```

2: $\int \text{Sin}\left[c+d\,(e+f\,x)^n\right] dx$ when $n-2\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\text{Sin}[z] = \frac{1}{2}\,i\,e^{-i\,z}-\frac{1}{2}\,i\,e^{i\,z}$

Basis: $\text{Cos}[z] = \frac{1}{2}\,e^{-i\,z}+\frac{1}{2}\,e^{i\,z}$

Rule: If $n-2\in\mathbb{Z}^+$, then

$$\int \text{Sin}\left[c+d\,(e+f\,x)^n\right] dx \longrightarrow \frac{i}{2}\int e^{-c\,i-d\,i\,(e+f\,x)^n}\,dx-\frac{i}{2}\int e^{c\,i+d\,i\,(e+f\,x)^n}\,dx$$

Program code:

```
Int[Sin[c_.+d_.*(e_.+f_.*x_)^n_],x_Symbol] :=
  I/2*Int[E^(-c*I-d*I*(e+f*x)^n),x] - I/2*Int[E^(c*I+d*I*(e+f*x)^n),x] /;
FreeQ[{c,d,e,f},x] && IGtQ[n,2]
```

```
Int[Cos[c_.+d_.*(e_.+f_.*x_)^n_],x_Symbol] :=
  1/2*Int[E^(-c*I-d*I*(e+f*x)^n),x] + 1/2*Int[E^(c*I+d*I*(e+f*x)^n),x] /;
FreeQ[{c,d,e,f},x] && IGtQ[n,2]
```

2: $\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n - 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p - 1 \in \mathbb{Z}^+ \land n - 1 \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx \longrightarrow \int \text{TrigReduce}[(a + b \operatorname{Sin}[c + d (e + f x)^n])^p, x] \, dx$$

Program code:

```
Int[(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Sin[c+d*(e+f*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,1] && IGtQ[n,1]
```

```
Int[(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Cos[c+d*(e+f*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,1] && IGtQ[n,1]
```

2: $\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx$ when $p \in \mathbb{Z}^+ \wedge n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $F[(e + f x)^n] = -\frac{1}{f} \operatorname{Subst}\left[\frac{F[x^{-n}]}{x^2}, x, \frac{1}{e+f x}\right] \partial_x \frac{1}{e+f x}$

Rule: If $p \in \mathbb{Z}^+ \wedge n \in \mathbb{Z}^-$, then

$$\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx \longrightarrow -\frac{1}{f} \operatorname{Subst}\left[\int \frac{(a + b \operatorname{Sin}[c + d x^{-n}])^p}{x^2} \, dx, x, \frac{1}{e + f x}\right]$$

Program code:

```
Int[(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  -1/f*Subst[Int[(a+b*Sin[c+d*x^(-n)])^p/x^2,x],x,1/(e+f*x)] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && ILtQ[n,0] && EqQ[n,-2]
```

```
Int[(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  -1/f*Subst[Int[(a+b*Cos[c+d*x^(-n)])^p/x^2,x],x,1/(e+f*x)] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && ILtQ[n,0] && EqQ[n,-2]
```

2: $\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx$ when $p \in \mathbb{Z}^+ \land \frac{1}{n} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $-1 \le n \le 1$, then $F[(e + f x)^n] = \frac{1}{n f} \operatorname{Subst}[x^{1/n-1} F[x], x, (e + f x)^n] \, \partial_x (e + f x)^n$

Rule: If $p \in \mathbb{Z}^+ \land \frac{1}{n} \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx \to \frac{1}{n f} \operatorname{Subst}\left[\int x^{1/n-1} (a + b \operatorname{Sin}[c + d x])^p \, dx, x, (e + f x)^n\right]$$

Program code:

```
Int[(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/(n*f)*Subst[Int[x^(1/n-1)*(a+b*Sin[c+d*x])^p,x],x,(e+f*x)^n] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IntegerQ[1/n]
```

```
Int[(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/(n*f)*Subst[Int[x^(1/n-1)*(a+b*Cos[c+d*x])^p,x],x,(e+f*x)^n] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IntegerQ[1/n]
```

3: $\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx$ when $p \in \mathbb{Z}^+ \land n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[(e + f x)^n] == \frac{k}{f} \text{Subst}[x^{k-1} F[x^{k n}], x, (e + f x)^{1/k}] \, \partial_x (e + f x)^{1/k}$

Rule: If $p \in \mathbb{Z}^+ \land n \in \mathbb{F}$, let $k = \text{Denominator}[n]$, then

$$\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx \longrightarrow \frac{k}{f} \text{Subst}\left[\int x^{k-1} (a + b \operatorname{Sin}[c + d x^{k n}])^p \, dx, x, (e + f x)^{1/k}\right]$$

Program code:

```
Int[(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k/f*Subst[Int[x^(k-1)*(a+b*Sin[c+d*x^(k*n)])^p,x],x,(e+f*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && FractionQ[n]
```

```
Int[(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k/f*Subst[Int[x^(k-1)*(a+b*Cos[c+d*x^(k*n)])^p,x],x,(e+f*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && FractionQ[n]
```

4. $\int (a + b \operatorname{Sin}[c + d (e + f x)^n])^p \, dx$ when $p \in \mathbb{Z}^+$

1: $\int \operatorname{Sin}[c + d (e + f x)^n] \, dx$

Derivation: Algebraic expansion

Basis: $\operatorname{Sin}[z] == \frac{1}{2} i e^{-i z} - \frac{1}{2} i e^{i z}$

Basis: $\operatorname{Cos}[z] == \frac{1}{2} e^{-i z} + \frac{1}{2} e^{i z}$

Rule:

$$\int \operatorname{Sin}[c + d (e + f x)^n] \, dx \longrightarrow \frac{i}{2} \int e^{-c i - d i (e+f x)^n} \, dx - \frac{i}{2} \int e^{c i + d i (e+f x)^n} \, dx$$

Program code:

```
Int[Sin[c_.+d_.*(e_.+f_.*x_)^n_],x_Symbol] :=
  I/2*Int[E^(-c*I-d*I*(e+f*x)^n),x] - I/2*Int[E^(c*I+d*I*(e+f*x)^n),x] /;
FreeQ[{c,d,e,f,n},x]
```

```
Int[Cos[c_.+d_.*(e_.+f_.*x_)^n_],x_Symbol] :=
  1/2*Int[E^(-c*I-d*I*(e+f*x)^n),x] + 1/2*Int[E^(c*I+d*I*(e+f*x)^n),x] /;
FreeQ[{c,d,e,f,n},x]
```

2: $\int (a + b\,\mathtt{Sin}[c + d\,(e + f\,x)^n])^p\,dx$ when $p - 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p - 1 \in \mathbb{Z}^+$, then

$$\int (a + b\,\mathtt{Sin}[c + d\,(e + f\,x)^n])^p\,dx \rightarrow \int \mathtt{TrigReduce}[(a + b\,\mathtt{Sin}[c + d\,(e + f\,x)^n])^p]\,dx$$

Program code:

```
Int[(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Sin[c+d*(e+f*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[p,1]
```

```
Int[(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Cos[c+d*(e+f*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[p,1]
```

X: $\int (a + b\,\mathtt{Sin}[c + d\,(e + f\,x)^n])^p\,dx$

Rule:

$$\int (a + b\,\mathtt{Sin}[c + d\,(e + f\,x)^n])^p\,dx \rightarrow \int (a + b\,\mathtt{Sin}[c + d\,(e + f\,x)^n])^p\,dx$$

Program code:

```
Int[(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_,x_Symbol] :=
  Unintegrable[(a+b*Sin[c+d*(e+f*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,n,p},x]
```

```
Int[(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_,x_Symbol] :=
  Unintegrable[(a+b*Cos[c+d*(e+f*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,n,p},x]
```

N. $\int (a + b\,\text{Sin}[u])^p\,dx$

1: $\int (a + b\,\text{Sin}[c + d\,u^n])^p\,dx$ when $u == e + f\,x$

Derivation: Algebraic normalization

Rule: If $u == e + f\,x$, then

$$\int (a + b\,\text{Sin}[c + d\,u^n])^p\,dx \longrightarrow \int (a + b\,\text{Sin}[c + d\,(e + f\,x)^n])^p\,dx$$

Program code:

```
Int[(a_.+b_.*Sin[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  Int[(a+b*Sin[c+d*ExpandToSum[u,x]^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && Not[LinearMatchQ[u,x]]
```

```
Int[(a_.+b_.*Cos[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  Int[(a+b*Cos[c+d*ExpandToSum[u,x]^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && Not[LinearMatchQ[u,x]]
```

2: $\int (a + b \operatorname{Sin}[u])^p \, dx$ when $u == c + d x^n$

Derivation: Algebraic normalization

Rule: If $u == c + d x^n$, then

$$\int (a + b \operatorname{Sin}[u])^p \, dx \to \int \left(a + b \operatorname{Sin}\left[c + d x^n\right]\right)^p \, dx$$

Program code:

```
Int[(a_.+b_.*Sin[u_])^p_.,x_Symbol] :=
  Int[(a+b*Sin[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(a_.+b_.*Cos[u_])^p_.,x_Symbol] :=
  Int[(a+b*Cos[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $(e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p$

1. $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

   1. $\int x^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

      1. $\int \frac{\text{Sin}[c+d\,x^n]}{x}\,dx$

         1: $\int \frac{\text{Sin}[d\,x^n]}{x}\,dx$

Derivation: Primitive rule

Basis: $\text{SinIntegral}'[z] == \frac{\text{Sin}[z]}{z}$

Rule:

$$\int \frac{\text{Sin}[d\,x^n]}{x}\,dx \to \frac{\text{SinIntegral}[d\,x^n]}{n}$$

Program code:

```
Int[Sin[d_.*x_^n_]/x_,x_Symbol] :=
  SinIntegral[d*x^n]/n /;
FreeQ[{d,n},x]
```

```
Int[Cos[d_.*x_^n_]/x_,x_Symbol] :=
  CosIntegral[d*x^n]/n /;
FreeQ[{d,n},x]
```

2: $\int \frac{\text{Sin}[c + d\, x^n]}{x}\, dx$

Derivation: Algebraic expansion

Basis: $\text{Sin}[w + z] == \text{Sin}[w]\, \text{Cos}[z] + \text{Cos}[w]\, \text{Sin}[z]$

Rule:

$$\int \frac{\text{Sin}[c + d\, x^n]}{x}\, dx \longrightarrow \text{Sin}[c] \int \frac{\text{Cos}[d\, x^n]}{x}\, dx + \text{Cos}[c] \int \frac{\text{Sin}[d\, x^n]}{x}\, dx$$

Program code:

```
Int[Sin[c_+d_.*x_^n_]/x_,x_Symbol] :=
  Sin[c]*Int[Cos[d*x^n]/x,x] + Cos[c]*Int[Sin[d*x^n]/x,x] /;
FreeQ[{c,d,n},x]
```

```
Int[Cos[c_+d_.*x_^n_]/x_,x_Symbol] :=
  Cos[c]*Int[Cos[d*x^n]/x,x] - Sin[c]*Int[Sin[d*x^n]/x,x] /;
FreeQ[{c,d,n},x]
```

2: $\int x^m \left(a + b \operatorname{Sin}[c + d x^n]\right)^p \, dx$ when $\frac{m+1}{n} \in \mathbb{Z} \land \left(p == 1 \lor m == n - 1 \lor p \in \mathbb{Z} \land \frac{m+1}{n} > 0\right)$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n} \in \mathbb{Z}$, then $x^m F[x^n] == \frac{1}{n} \operatorname{Subst}\left[x^{\frac{m+1}{n}-1} F[x], x, x^n\right] \partial_x x^n$

Rule: If $\frac{m+1}{n} \in \mathbb{Z} \land \left(p == 1 \lor m == n - 1 \lor p \in \mathbb{Z} \land \frac{m+1}{n} > 0\right)$, then

$$\int x^m \left(a + b \operatorname{Sin}[c + d x^n]\right)^p dx \to \frac{1}{n} \operatorname{Subst}\left[\int x^{\frac{m+1}{n}-1} \left(a + b \operatorname{Sin}[c + d x]\right)^p dx, x, x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Sin[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]] && (EqQ[p,1] || EqQ[m,n-1] || IntegerQ[p] && GtQ[Simplify[(m+1)/n],0])
```

```
Int[x_^m_.*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Cos[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]] && (EqQ[p,1] || EqQ[m,n-1] || IntegerQ[p] && GtQ[Simplify[(m+1)/n],0])
```

2: $\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Rule: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx \to \frac{e^{\mathrm{IntPart}[m]}\,(e\,x)^{\mathrm{FracPart}[m]}}{x^{\mathrm{FracPart}[m]}}\int x^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$$

Program code:

```
Int[(e_*x_)^m_*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sin[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[(e_*x_)^m_*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Cos[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]]
```

2. $\int (e\,x)^m \left(a+b\,\mathrm{Sin}\left[c+d\,x^n\right]\right)^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}$

1. $\int (e\,x)^m \left(a+b\,\mathrm{Sin}\left[c+d\,x^n\right]\right)^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}^+$

1. $\int (e\,x)^m\,\mathrm{Sin}\left[c+d\,x^n\right]\,dx$

1: $\int x^{\frac{n}{2}-1}\,\mathrm{Sin}\left[a+b\,x^n\right]\,dx$

Derivation: Integration by substitution

Basis: $x^{\frac{n}{2}-1}\,F\left[x^n\right] == \frac{2}{n}\,\mathrm{Subst}\left[F\left[x^2\right],\,x,\,x^{\frac{n}{2}}\right]\,\partial_x x^{\frac{n}{2}}$

Note: Although not essential, this rule produces antiderivatives in terms of Fresnel integrals instead of complex error functions.

Rule:

$$\int x^{\frac{n}{2}-1}\,\mathrm{Sin}\left[a+b\,x^n\right]\,dx \longrightarrow \frac{2}{n}\,\mathrm{Subst}\left[\int \mathrm{Sin}\left[a+b\,x^2\right]\,dx,\,x,\,x^{\frac{n}{2}}\right]$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_],x_Symbol] :=
  2/n*Subst[Int[Sin[a+b*x^2],x],x,x^(n/2)] /;
FreeQ[{a,b,m,n},x] && EqQ[m,n/2-1]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_],x_Symbol] :=
  2/n*Subst[Int[Cos[a+b*x^2],x],x,x^(n/2)] /;
FreeQ[{a,b,m,n},x] && EqQ[m,n/2-1]
```

2: $\int (e\,x)^m \,\mathrm{Sin}[c+d\,x^n]\,dx$ when $n\in\mathbb{Z}^+\wedge 0<n<m+1$

Reference: CRC 392, A&S 4.3.119

Reference: CRC 396, A&S 4.3.123

Derivation: Integration by parts

Basis: If $n\in\mathbb{Z}$, then $(e\,x)^m\,\mathrm{Sin}[c+d\,x^n] == -\frac{e^{n-1}\,(e\,x)^{m-n+1}}{d\,n}\,\partial_x\,\mathrm{Cos}[c+d\,x^n]$

Rule: If $n\in\mathbb{Z}^+\wedge 0<n<m+1$, then

$$\int (e\,x)^m\,\mathrm{Sin}[c+d\,x^n]\,dx \longrightarrow -\frac{e^{n-1}\,(e\,x)^{m-n+1}\,\mathrm{Cos}[c+d\,x^n]}{d\,n}+\frac{e^n\,(m-n+1)}{d\,n}\int (e\,x)^{m-n}\,\mathrm{Cos}[c+d\,x^n]\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[c_.+d_.*x_^n_],x_Symbol] :=
  -e^(n-1)*(e*x)^(m-n+1)*Cos[c+d*x^n]/(d*n) +
  e^n*(m-n+1)/(d*n)*Int[(e*x)^(m-n)*Cos[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[n,m+1]
```

```
Int[(e_.*x_)^m_.*Cos[c_.+d_.*x_^n_],x_Symbol] :=
  e^(n-1)*(e*x)^(m-n+1)*Sin[c+d*x^n]/(d*n) -
  e^n*(m-n+1)/(d*n)*Int[(e*x)^(m-n)*Sin[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[n,m+1]
```

3: $\int (e\,x)^m \,\mathrm{Sin}\left[c + d\,x^n\right] \,dx$ when $n \in \mathbb{Z}^+ \land m < -1$

Reference: CRC 405, A&S 4.3.120

Reference: CRC 406, A&S 4.3.124

Derivation: Integration by parts

Rule: If $n \in \mathbb{Z}^+ \land m < -1$, then

$$\int (e\,x)^m \,\mathrm{Sin}\left[c + d\,x^n\right] \,dx \longrightarrow \frac{(e\,x)^{m+1}\,\mathrm{Sin}\left[c + d\,x^n\right]}{e\,(m+1)} - \frac{d\,n}{e^n\,(m+1)} \int (e\,x)^{m+n}\,\mathrm{Cos}\left[c + d\,x^n\right] \,dx$$

Program code:

```
Int[(e_.*x_)^m_*Sin[c_.+d_.*x_^n_],x_Symbol] :=
  (e*x)^(m+1)*Sin[c+d*x^n]/(e*(m+1)) -
  d*n/(e^n*(m+1))*Int[(e*x)^(m+n)*Cos[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[m,-1]
```

```
Int[(e_.*x_)^m_*Cos[c_.+d_.*x_^n_],x_Symbol] :=
  (e*x)^(m+1)*Cos[c+d*x^n]/(e*(m+1)) +
  d*n/(e^n*(m+1))*Int[(e*x)^(m+n)*Sin[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[m,-1]
```

4: $\int (e\,x)^m \operatorname{Sin}[c+d\,x^n]\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\operatorname{Sin}[z] == \frac{1}{2}\,i\,e^{-i\,z} - \frac{1}{2}\,i\,e^{i\,z}$

Basis: $\operatorname{Cos}[z] == \frac{1}{2}\,e^{-i\,z} + \frac{1}{2}\,e^{i\,z}$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int (e\,x)^m \operatorname{Sin}[c+d\,x^n]\,dx \longrightarrow \frac{i}{2}\int (e\,x)^m e^{-c\,i-d\,i\,x^n}\,dx - \frac{i}{2}\int (e\,x)^m e^{c\,i+d\,i\,x^n}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[c_.+d_.*x_^n_],x_Symbol] :=
  I/2*Int[(e*x)^m*E^(-c*I-d*I*x^n),x] - I/2*Int[(e*x)^m*E^(c*I+d*I*x^n),x] /;
FreeQ[{c,d,e,m},x] && IGtQ[n,0]
```

```
Int[(e_.*x_)^m_.*Cos[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(-c*I-d*I*x^n),x] + 1/2*Int[(e*x)^m*E^(c*I+d*I*x^n),x] /;
FreeQ[{c,d,e,m},x] && IGtQ[n,0]
```

2. $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p>1$

0: $\int x^m\,\text{Sin}[a+b\,x^n]^2\,dx$

Derivation: Algebraic expansion

Basis: $\text{Sin}[z]^2 == \frac{1}{2}-\frac{\text{Cos}[2\,z]}{2}$

Rule:

$$\int x^m\,\text{Sin}[a+b\,x^n]^2\,dx \longrightarrow \frac{1}{2}\int x^m\,dx-\frac{1}{2}\int x^m\,\text{Cos}[2\,a+2\,b\,x^n]\,dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_/2]^2,x_Symbol] :=
  1/2*Int[x^m,x] - 1/2*Int[x^m*Cos[2*a+b*x^n],x] /;
FreeQ[{a,b,m,n},x]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_/2]^2,x_Symbol] :=
  1/2*Int[x^m,x] + 1/2*Int[x^m*Cos[2*a+b*x^n],x] /;
FreeQ[{a,b,m,n},x]
```

1: $\int x^m \operatorname{Sin}\left[a+b\,x^n\right]^p \, dx$ when $p-1\in\mathbb{Z}^+ \land m+n=0 \land n\neq 1 \land n\in\mathbb{Z}$

Derivation: Integration by parts

Rule: If $p-1\in\mathbb{Z}^+ \land m+n=0 \land n\neq 1 \land n\in\mathbb{Z}$, then

$$\int x^m \operatorname{Sin}\left[a+b\,x^n\right]^p \, dx \longrightarrow \frac{x^{m+1}\operatorname{Sin}\left[a+b\,x^n\right]^p}{m+1} - \frac{b\,n\,p}{m+1}\int \operatorname{Sin}\left[a+b\,x^n\right]^{p-1}\operatorname{Cos}\left[a+b\,x^n\right] dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m+1)*Sin[a+b*x^n]^p/(m+1) -
  b*n*p/(m+1)*Int[Sin[a+b*x^n]^(p-1)*Cos[a+b*x^n],x] /;
FreeQ[{a,b},x] && IGtQ[p,1] && EqQ[m+n,0] && NeQ[n,1] && IntegerQ[n]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m+1)*Cos[a+b*x^n]^p/(m+1) +
  b*n*p/(m+1)*Int[Cos[a+b*x^n]^(p-1)*Sin[a+b*x^n],x] /;
FreeQ[{a,b},x] && IGtQ[p,1] && EqQ[m+n,0] && NeQ[n,1] && IntegerQ[n]
```

2: $\int x^m \operatorname{Sin}[a+b\,x^n]^p\,dx$ when $m-2n+1==0\,\land\,p>1$

Reference: G&R 2.631.2' special case when $m-2n+1==0$

Reference: G&R 2.631.3' special case when $m-2n+1==0$

Rule: If $m-2n+1==0\,\land\,p>1$, then

$$\int x^m \operatorname{Sin}[a+b\,x^n]^p\,dx \longrightarrow \frac{n \operatorname{Sin}[a+b\,x^n]^p}{b^2\,n^2\,p^2}-\frac{x^n \operatorname{Cos}[a+b\,x^n]\operatorname{Sin}[a+b\,x^n]^{p-1}}{b\,n\,p}+\frac{p-1}{p}\int x^m \operatorname{Sin}[a+b\,x^n]^{p-2}\,dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  n*Sin[a+b*x^n]^p/(b^2*n^2*p^2) -
  x^n*Cos[a+b*x^n]*Sin[a+b*x^n]^(p-1)/(b*n*p) +
  (p-1)/p*Int[x^m*Sin[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1,0] && GtQ[p,1]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  n*Cos[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^n*Sin[a+b*x^n]*Cos[a+b*x^n]^(p-1)/(b*n*p) +
  (p-1)/p*Int[x^m*Cos[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1,0] && GtQ[p,1]
```

3: $\int x^m \operatorname{Sin}[a+b x^n]^p \, dx$ when $p>1 \land n \in \mathbb{Z}^+ \land m-2n+1 \in \mathbb{Z}^+$

Reference: G&R 2.631.2'

Reference: G&R 2.631.3'

Rule: If $p>1 \land n \in \mathbb{Z}^+ \land m-2n+1 \in \mathbb{Z}^+$, then

$$\int x^m \operatorname{Sin}[a+b x^n]^p \, dx \longrightarrow$$
$$\frac{(m-n+1)\, x^{m-2n+1} \operatorname{Sin}[a+b x^n]^p}{b^2 n^2 p^2} - \frac{x^{m-n+1} \operatorname{Cos}[a+b x^n] \operatorname{Sin}[a+b x^n]^{p-1}}{b\, n\, p} + \frac{p-1}{p} \int x^m \operatorname{Sin}[a+b x^n]^{p-2} \, dx - \frac{(m-n+1)\,(m-2n+1)}{b^2 n^2 p^2} \int x^{m-2n} \operatorname{Sin}[a+b x^n]^p \, dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  (m-n+1)*x^(m-2*n+1)*Sin[a+b*x^n]^p/(b^2*n^2*p^2) -
  x^(m-n+1)*Cos[a+b*x^n]*Sin[a+b*x^n]^(p-1)/(b*n*p) +
  (p-1)/p*Int[x^m*Sin[a+b*x^n]^(p-2),x] -
  (m-n+1)*(m-2*n+1)/(b^2*n^2*p^2)*Int[x^(m-2*n)*Sin[a+b*x^n]^p,x] /;
FreeQ[{a,b},x] && GtQ[p,1] && IGtQ[n,0] && IGtQ[m,2*n-1]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  (m-n+1)*x^(m-2*n+1)*Cos[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^(m-n+1)*Sin[a+b*x^n]*Cos[a+b*x^n]^(p-1)/(b*n*p) +
  (p-1)/p*Int[x^m*Cos[a+b*x^n]^(p-2),x] -
  (m-n+1)*(m-2*n+1)/(b^2*n^2*p^2)*Int[x^(m-2*n)*Cos[a+b*x^n]^p,x] /;
FreeQ[{a,b},x] && GtQ[p,1] && IGtQ[n,0] && IGtQ[m,2*n-1]
```

4: $\int x^m \operatorname{Sin}\left[a+b x^n\right]^p \, dx$ when $p>1 \land n \in \mathbb{Z}^+ \land m+2n-1 \in \mathbb{Z}^- \land m+n+1 \neq 0$

Reference: G&R 2.638.1'

Reference: G&R 2.638.2'

Rule: If $p>1 \land n \in \mathbb{Z}^+ \land m+2n-1 \in \mathbb{Z}^- \land m+n+1 \neq 0$, then

$$\int x^m \operatorname{Sin}\left[a+b x^n\right]^p \, dx \longrightarrow$$

$$\frac{x^{m+1} \operatorname{Sin}\left[a+b x^n\right]^p}{m+1}-\frac{b n p x^{m+n+1} \operatorname{Cos}\left[a+b x^n\right] \operatorname{Sin}\left[a+b x^n\right]^{p-1}}{(m+1)(m+n+1)}-$$
$$\frac{b^2 n^2 p^2}{(m+1)(m+n+1)} \int x^{m+2n} \operatorname{Sin}\left[a+b x^n\right]^p \, dx+\frac{b^2 n^2 p(p-1)}{(m+1)(m+n+1)} \int x^{m+2n} \operatorname{Sin}\left[a+b x^n\right]^{p-2} \, dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m+1)*Sin[a+b*x^n]^p/(m+1) -
  b*n*p*x^(m+n+1)*Cos[a+b*x^n]*Sin[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) -
  b^2*n^2*p^2/((m+1)*(m+n+1))*Int[x^(m+2*n)*Sin[a+b*x^n]^p,x] +
  b^2*n^2*p*(p-1)/((m+1)*(m+n+1))*Int[x^(m+2*n)*Sin[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && GtQ[p,1] && IGtQ[n,0] && ILtQ[m,-2*n+1] && NeQ[m+n+1,0]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m+1)*Cos[a+b*x^n]^p/(m+1) +
  b*n*p*x^(m+n+1)*Sin[a+b*x^n]*Cos[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) -
  b^2*n^2*p^2/((m+1)*(m+n+1))*Int[x^(m+2*n)*Cos[a+b*x^n]^p,x] +
  b^2*n^2*p*(p-1)/((m+1)*(m+n+1))*Int[x^(m+2*n)*Cos[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && GtQ[p,1] && IGtQ[n,0] && ILtQ[m,-2*n+1] && NeQ[m+n+1,0]
```

5: $\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^+\land\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $(e\,x)^m\,F[x]=\frac{k}{e}\,\mathrm{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{e}\right],\,x,\,(e\,x)^{1/k}\right]\,\partial_x\,(e\,x)^{1/k}$

Rule: If $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^+\land\,m\in\mathbb{F}$, let $k=\mathrm{Denominator}[m]$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx\,\longrightarrow\,\frac{k}{e}\,\mathrm{Subst}\left[\int x^{k\,(m+1)-1}\left(a+b\,\mathrm{Sin}\left[c+\frac{d\,x^{k\,n}}{e^n}\right]\right)^p dx,\,x,\,(e\,x)^{1/k}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  k/e*Subst[Int[x^(k*(m+1)-1)*(a+b*Sin[c+d*x^(k*n)/e^n])^p,x],x,(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p] && IGtQ[n,0] && FractionQ[m]
```

```
Int[(e_.*x_)^m_*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  k/e*Subst[Int[x^(k*(m+1)-1)*(a+b*Cos[c+d*x^(k*n)/e^n])^p,x],x,(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p] && IGtQ[n,0] && FractionQ[m]
```

6: $\int (e\,x)^m\,\left(a+b\,\text{Sin}\left[c+d\,x^n\right]\right)^p\,dx$ when $p-1\in\mathbb{Z}^+\land\,n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p-1\in\mathbb{Z}^+\land\,n\in\mathbb{Z}^+$, then

$$\int (e\,x)^m\,\left(a+b\,\text{Sin}\left[c+d\,x^n\right]\right)^p\,dx\,\longrightarrow\,\int (e\,x)^m\,\text{TrigReduce}\left[\left(a+b\,\text{Sin}\left[c+d\,x^n\right]\right)^p,\,x\right]\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Sin[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[p,1] && IGtQ[n,0]
```

```
Int[(e_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Cos[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[p,1] && IGtQ[n,0]
```

3. $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p<-1$

1: $\int x^m\,\text{Sin}[a+b\,x^n]^p\,dx$ when $m-2\,n+1==0\,\land\,p<-1\,\land\,p\neq-2$

Reference: G&R 2.643.1' special case when $m-2\,n+1==0$

Reference: G&R 2.643.2' special case when $m-2\,n+1==0$

Rule: If $m-2\,n+1==0\,\land\,p<-1\,\land\,p\neq-2$, then

$$\int x^m\,\text{Sin}[a+b\,x^n]^p\,dx \longrightarrow \frac{x^n\,\text{Cos}[a+b\,x^n]\,\text{Sin}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)} - \frac{n\,\text{Sin}[a+b\,x^n]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)} + \frac{p+2}{p+1}\int x^m\,\text{Sin}[a+b\,x^n]^{p+2}\,dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^n*Cos[a+b*x^n]*Sin[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  n*Sin[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  (p+2)/(p+1)*Int[x^m*Sin[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1,0] && LtQ[p,-1] && NeQ[p,-2]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -x^n*Sin[a+b*x^n]*Cos[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  n*Cos[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  (p+2)/(p+1)*Int[x^m*Cos[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1,0] && LtQ[p,-1] && NeQ[p,-2]
```

2: $\int x^m \operatorname{Sin}\left[a+b\,x^n\right]^p \, dx$ when $(m \mid n) \in \mathbb{Z} \land p < -1 \land p \neq -2 \land 0 < 2\,n < m+1$

Reference: G&R 2.643.1'

Reference: G&R 2.643.2

Rule: If $(m \mid n) \in \mathbb{Z} \land p < -1 \land p \neq -2 \land 0 < 2\,n < m+1$, then

$$\int x^m \operatorname{Sin}\left[a+b\,x^n\right]^p \, dx \longrightarrow$$

$$\frac{x^{m-n+1} \operatorname{Cos}\left[a+b\,x^n\right] \operatorname{Sin}\left[a+b\,x^n\right]^{p+1}}{b\,n\,(p+1)} - \frac{(m-n+1)\,x^{m-2\,n+1} \operatorname{Sin}\left[a+b\,x^n\right]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)} +$$

$$\frac{p+2}{p+1} \int x^m \operatorname{Sin}\left[a+b\,x^n\right]^{p+2} dx + \frac{(m-n+1)\,(m-2\,n+1)}{b^2\,n^2\,(p+1)\,(p+2)} \int x^{m-2\,n} \operatorname{Sin}\left[a+b\,x^n\right]^{p+2} dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m-n+1)*Cos[a+b*x^n]*Sin[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)*x^(m-2*n+1)*Sin[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  (p+2)/(p+1)*Int[x^m*Sin[a+b*x^n]^(p+2),x] +
  (m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2))*Int[x^(m-2*n)*Sin[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b},x] && LtQ[p,-1] && NeQ[p,-2] && IGtQ[n,0] && IGtQ[m,2*n-1]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -x^(m-n+1)*Sin[a+b*x^n]*Cos[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)*x^(m-2*n+1)*Cos[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  (p+2)/(p+1)*Int[x^m*Cos[a+b*x^n]^(p+2),x] +
  (m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2))*Int[x^(m-2*n)*Cos[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b},x] && LtQ[p,-1] && NeQ[p,-2] && IGtQ[n,0] && IGtQ[m,2*n-1]
```

2. $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+\land n\in\mathbb{Z}^-$

   1. $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+\land n\in\mathbb{Z}^-\land m\in\mathbb{Q}$

      1: $\int x^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+\land n\in\mathbb{Z}^-\land m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\land m\in\mathbb{Z}$, then $x^m\,F[x^n] = -\text{Subst}\left[\frac{F[x^{-n}]}{x^{m+2}},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule: If $p\in\mathbb{Z}^+\land n\in\mathbb{Z}^-\land m\in\mathbb{Z}$, then

$$\int x^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx\;\longrightarrow\;-\text{Subst}\left[\int\frac{(a+b\,\text{Sin}[c+d\,x^{-n}])^p}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(a+b*Sin[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0] && ILtQ[n,0] && IntegerQ[m] && EqQ[n,-2]
```

```
Int[x_^m_.*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(a+b*Cos[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0] && ILtQ[n,0] && IntegerQ[m] && EqQ[n,-2]
```

2: $\int (e\,x)^m\,\left(a+b\,\mathrm{Sin}\left[c+d\,x^n\right]\right)^p\,dx$ when $p\in\mathbb{Z}^+\land n\in\mathbb{Z}^-\land m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\ \land\ k>1$, then $(e\,x)^m\,F[x^n]=-\frac{k}{e}\,\mathrm{Subst}\left[\frac{F\left[e^{-n}\,x^{-k\,n}\right]}{x^{k\,(m+1)+1}},\,x,\,\frac{1}{(e\,x)^{1/k}}\right]\,\partial_x\,\frac{1}{(e\,x)^{1/k}}$

Rule: If $p\in\mathbb{Z}^+\land\ n\in\mathbb{Z}^-\land\ m\in\mathbb{F}$, let $k=\mathrm{Denominator}[m]$, then

$$\int (e\,x)^m\,\left(a+b\,\mathrm{Sin}\left[c+d\,x^n\right]\right)^p\,dx\ \longrightarrow\ -\frac{k}{e}\,\mathrm{Subst}\left[\int\frac{\left(a+b\,\mathrm{Sin}\left[c+d\,e^{-n}\,x^{-k\,n}\right]\right)^p}{x^{k\,(m+1)+1}}\,dx,\,x,\,\frac{1}{(e\,x)^{1/k}}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/e*Subst[Int[(a+b*Sin[c+d/(e^n*x^(k*n))])^p/x^(k*(m+1)+1),x],x,1/(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && ILtQ[n,0] && FractionQ[m]
```

```
Int[(e_.*x_)^m_*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/e*Subst[Int[(a+b*Cos[c+d/(e^n*x^(k*n))])^p/x^(k*(m+1)+1),x],x,1/(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && ILtQ[n,0] && FractionQ[m]
```

2: $\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+\land n\in\mathbb{Z}^-\land m\notin\mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\left((e\,x)^m\,\left(x^{-1}\right)^m\right)=0$

Basis: $F[x]=-\mathrm{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule: If $p\in\mathbb{Z}^+\land n\in\mathbb{Z}^-\land m\notin\mathbb{Q}$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx\;\rightarrow\;(e\,x)^m\,\left(x^{-1}\right)^m\int\frac{(a+b\,\mathrm{Sin}[c+d\,x^n])^p}{\left(x^{-1}\right)^m}\,dx\;\rightarrow\;-(e\,x)^m\,\left(x^{-1}\right)^m\,\mathrm{Subst}\left[\int\frac{(a+b\,\mathrm{Sin}[c+d\,x^{-n}])^p}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -(e*x)^m*(x^(-1))^m*Subst[Int[(a+b*Sin[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[p,0] && ILtQ[n,0] && Not[RationalQ[m]]
```

```
Int[(e_.*x_)^m_*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -(e*x)^m*(x^(-1))^m*Subst[Int[(a+b*Cos[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[p,0] && ILtQ[n,0] && Not[RationalQ[m]]
```

3. $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p \in \mathbb{Z} \land n \in \mathbb{F}$

1: $\int x^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p \in \mathbb{Z} \land n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $x^m\,F[x^n] = k\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[x^{k\,n}\right],\,x,\,x^{1/k}\right]\,\partial_x\,x^{1/k}$

Rule: If $p \in \mathbb{Z} \land n \in \mathbb{F}$, let $k = \text{Denominator}[n]$, then

$$\int x^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx \longrightarrow k\,\text{Subst}\left[\int x^{k\,(m+1)-1}\,\left(a+b\,\text{Sin}\left[c+d\,x^{k\,n}\right]\right)^p\,dx,\,x,\,x^{1/k}\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*Sin[c+d*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,m},x] && IntegerQ[p] && FractionQ[n]
```

```
Int[x_^m_.*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*Cos[c+d*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,m},x] && IntegerQ[p] && FractionQ[n]
```

2: $\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Rule: If $p\in\mathbb{Z}\,\land\,n\in\mathbb{F}$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx \longrightarrow \frac{e^{\mathrm{IntPart}[m]}\,(e\,x)^{\mathrm{FracPart}[m]}}{x^{\mathrm{FracPart}[m]}}\int x^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$$

Program code:

```
Int[(e_*x_)^m_*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sin[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && IntegerQ[p] && FractionQ[n]
```

```
Int[(e_*x_)^m_*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Cos[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && IntegerQ[p] && FractionQ[n]
```

4. $\int (e\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\land m\neq -1\land \frac{n}{m+1}\in\mathbb{Z}^+$

1: $\int x^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\land m\neq -1\land \frac{n}{m+1}\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1}\in\mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{m+1}\,\mathrm{Subst}\left[F\left[x^{\frac{n}{m+1}}\right],\,x,\,x^{m+1}\right]\,\partial_x x^{m+1}$

Rule: If $p\in\mathbb{Z}\land m\neq -1\land \frac{n}{m+1}\in\mathbb{Z}^+$, then

$$\int x^m\,(a+b\,\mathrm{Sin}[c+d\,x^n])^p\,dx\;\longrightarrow\;\frac{1}{m+1}\,\mathrm{Subst}\left[\int\left(a+b\,\mathrm{Sin}\left[c+d\,x^{\frac{n}{m+1}}\right]\right)^p dx,\,x,\,x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*Sin[c+d*x^Simplify[n/(m+1)]])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

```
Int[x_^m_.*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*Cos[c+d*x^Simplify[n/(m+1)]])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

2: $\int (e\,x)^m\,\left(a+b\,\text{Sin}\left[c+d\,x^n\right]\right)^p\,dx$ when $p\in\mathbb{Z}\,\land\,m\neq-1\,\land\,\frac{n}{m+1}\in\mathbb{Z}^+$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Rule: If $p\in\mathbb{Z}\,\land\,m\neq-1\,\land\,\frac{n}{m+1}\in\mathbb{Z}^+$, then

$$\int (e\,x)^m\,\left(a+b\,\text{Sin}\left[c+d\,x^n\right]\right)^p\,dx\,\longrightarrow\,\frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,\left(a+b\,\text{Sin}\left[c+d\,x^n\right]\right)^p\,dx$$

Program code:

```
Int[(e_*x_)^m_*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sin[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

```
Int[(e_*x_)^m_*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Cos[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

5. $\int (e\,x)^m \left(a+b\,\mathrm{Sin}\left[c+d\,x^n\right]\right)^p\,dx$ when $p \in \mathbb{Z}^+$

1: $\int (e\,x)^m\,\mathrm{Sin}\left[c+d\,x^n\right]\,dx$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[z] == \frac{1}{2}\,i\,e^{-i\,z} - \frac{1}{2}\,i\,e^{i\,z}$

Basis: $\mathrm{Cos}[z] == \frac{1}{2}\,e^{-i\,z} + \frac{1}{2}\,e^{i\,z}$

Rule:

$$\int (e\,x)^m\,\mathrm{Sin}\left[c+d\,x^n\right]\,dx \longrightarrow \frac{i}{2}\int (e\,x)^m\,e^{-c\,i-d\,i\,x^n}\,dx - \frac{i}{2}\int (e\,x)^m\,e^{c\,i+d\,i\,x^n}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[c_.+d_.*x_^n_],x_Symbol] :=
  I/2*Int[(e*x)^m*E^(-c*I-d*I*x^n),x] - I/2*Int[(e*x)^m*E^(c*I+d*I*x^n),x] /;
FreeQ[{c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Cos[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(-c*I-d*I*x^n),x] + 1/2*Int[(e*x)^m*E^(c*I+d*I*x^n),x] /;
FreeQ[{c,d,e,m,n},x]
```

2: $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx \rightarrow \int (e\,x)^m\,\text{TrigReduce}[(a+b\,\text{Sin}[c+d\,x^n])^p,\,x]\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Sin[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

```
Int[(e_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Cos[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

X: $\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$

Rule:

$$\int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx \rightarrow \int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*Sin[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(e_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*Cos[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

N: $\int (e\,x)^m\,(a+b\,\text{Sin}[u])^p\,dx$ when $u == c+d\,x^n$

Derivation: Algebraic normalization

Rule: If $u == c+d\,x^n$, then

$$\int (e\,x)^m\,(a+b\,\text{Sin}[u])^p\,dx \longrightarrow \int (e\,x)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx$$

Program code:

```
Int[(e_*x_)^m_.*(a_.+b_.*Sin[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Sin[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(e_*x_)^m_.*(a_.+b_.*Cos[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Cos[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $(g+h\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,(e+f\,x)^n])^p$

1: $\int (g+h\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,(e+f\,x)^n])^p\,dx$ when $p\in\mathbb{Z}^+\wedge\frac{1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $-1\le n\le 1$, then $(g+h\,x)^m\,F[(e+f\,x)^n]=\frac{1}{n\,f}\,\mathrm{Subst}\left[x^{1/n-1}\left(g-\frac{e\,h}{f}+\frac{h\,x^{1/n}}{f}\right)^m F[x],\,x,\,(e+f\,x)^n\right]\partial_x(e+f\,x)^n$

Rule: If $p\in\mathbb{Z}^+\wedge\frac{1}{n}\in\mathbb{Z}$, then

$$\int (g+h\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,(e+f\,x)^n])^p\,dx\,\longrightarrow$$
$$\frac{1}{n\,f}\,\mathrm{Subst}\left[\int (a+b\,\mathrm{Sin}[c+d\,x])^p\,\mathrm{ExpandIntegrand}\left[x^{1/n-1}\left(g-\frac{e\,h}{f}+\frac{h\,x^{1/n}}{f}\right)^m,\,x\right]dx,\,x,\,(e+f\,x)^n\right]$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/(n*f)*Subst[Int[ExpandIntegrand[(a+b*Sin[c+d*x])^p,x^(1/n-1)*(g-e*h/f+h*x^(1/n)/f)^m,x],x],x,(e+f*x)^n] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IGtQ[p,0] && IntegerQ[1/n]
```

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/(n*f)*Subst[Int[ExpandIntegrand[(a+b*Cos[c+d*x])^p,x^(1/n-1)*(g-e*h/f+h*x^(1/n)/f)^m,x],x],x,(e+f*x)^n] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IGtQ[p,0] && IntegerQ[1/n]
```

x: $\int (g+h\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,(e+f\,x)^n])^p\,dx$ when $p\in\mathbb{Z}^+\land m\in\mathbb{Z}\land\frac{1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $m\in\mathbb{Z}\land\frac{1}{n}\in\mathbb{Z}$, then $(g+h\,x)^m\,F[(e+f\,x)^n]=\frac{1}{n\,f^{m+1}}\,\mathrm{Subst}[x^{1/n-1}\,(f\,g-e\,h+h\,x^{1/n})^m\,F[x],\,x,\,(e+f\,x)^n]\,\partial_x(e+f\,x)^n$

Rule: If $p\in\mathbb{Z}^+\land m\in\mathbb{Z}\land\frac{1}{n}\in\mathbb{Z}$, then

$$\int (g+h\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,(e+f\,x)^n])^p\,dx\,\longrightarrow$$
$$\frac{1}{n\,f^{m+1}}\,\mathrm{Subst}\Big[\int (a+b\,\mathrm{Sin}[c+d\,x])^p\,\mathrm{ExpandIntegrand}[x^{1/n-1}\,(f\,g-e\,h+h\,x^{1/n})^m,\,x]\,dx,\,x,\,(e+f\,x)^n\Big]$$

Program code:

```
(* Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/(n*f^(m+1))*Subst[Int[ExpandIntegrand[(a+b*Sin[c+d*x])^p,x^(1/n-1)*(f*g-e*h+h*x^(1/n))^m,x],x],x,(e+f*x)^n] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[p,0] && IntegerQ[m] && IntegerQ[1/n] *)
```

```
(* Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/(n*f^(m+1))*Subst[Int[ExpandIntegrand[(a+b*Cos[c+d*x])^p,x^(1/n-1)*(f*g-e*h+h*x^(1/n))^m,x],x],x,(e+f*x)^n] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[p,0] && IntegerQ[m] && IntegerQ[1/n] *)
```

2: $\int (g+h\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,(e+f\,x)^n])^p\,dx$ when $p\in\mathbb{Z}^+\land m\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $m\in\mathbb{Z}\land k\in\mathbb{Z}^+$, then $(g+h\,x)^m\,F[(e+f\,x)^n]=\frac{k}{f^{m+1}}\,\mathrm{Subst}[x^{k-1}\,(f\,g-e\,h+h\,x^k)^m\,F[x^{k\,n}],\,x,\,(e+f\,x)^{1/k}]\,\partial_x(e+f\,x)^{1/k}$

Rule: If $p\in\mathbb{Z}^+\land m\in\mathbb{Z}^+$, let $k=\mathrm{Denominator}[n]$, then

$$\int (g+h\,x)^m\,(a+b\,\mathrm{Sin}[c+d\,(e+f\,x)^n])^p\,dx\,\longrightarrow$$

$$\frac{k}{f^{m+1}}\,\text{Subst}\Big[\int (a+b\,\text{Sin}[c+d\,x^{k\,n}])^p\,\text{ExpandIntegrand}\big[x^{k-1}\,(f\,g-e\,h+h\,x^k)^m,\,x\big]\,dx,\,x,\,(e+f\,x)^{1/k}\Big]$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  Module[{k=If[FractionQ[n],Denominator[n],1]},
  k/f^(m+1)*Subst[Int[ExpandIntegrand[(a+b*Sin[c+d*x^(k*n)])^p,x^(k-1)*(f*g-e*h+h*x^k)^m,x],x],x,(e+f*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[p,0] && IGtQ[m,0]
```

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  Module[{k=If[FractionQ[n],Denominator[n],1]},
  k/f^(m+1)*Subst[Int[ExpandIntegrand[(a+b*Cos[c+d*x^(k*n)])^p,x^(k-1)*(f*g-e*h+h*x^k)^m,x],x],x,(e+f*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[p,0] && IGtQ[m,0]
```

3: $\int (g+h\,x)^m\,(a+b\,\text{Sin}[c+d\,(e+f\,x)^n])^p\,dx$ when $p\in\mathbb{Z}^+\wedge\,f\,g-e\,h==0$

Derivation: Integration by substitution

Basis: If $f\,g-e\,h==0$, then $(g+h\,x)^m\,F[e+f\,x]==\frac{1}{f}\,\text{Subst}\big[\left(\frac{h\,x}{f}\right)^m F[x],\,x,\,e+f\,x\big]\,\partial_x\,(e+f\,x)$

Note: If $p\in\mathbb{Z}^+$, then $\left(\frac{h\,x}{f}\right)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p$ is integrable wrt x.

Rule: If $p\in\mathbb{Z}^+\wedge\,f\,g-e\,h==0$, then

$$\int (g+h\,x)^m\,(a+b\,\text{Sin}[c+d\,(e+f\,x)^n])^p\,dx\;\rightarrow\;\frac{1}{f}\,\text{Subst}\Big[\int\left(\frac{h\,x}{f}\right)^m\,(a+b\,\text{Sin}[c+d\,x^n])^p\,dx,\,x,\,e+f\,x\Big]$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/f*Subst[Int[(h*x/f)^m*(a+b*Sin[c+d*x^n])^p,x],x,e+f*x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IGtQ[p,0] && EqQ[f*g-e*h,0]
```

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  1/f*Subst[Int[(h*x/f)^m*(a+b*Cos[c+d*x^n])^p,x],x,e+f*x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IGtQ[p,0] && EqQ[f*g-e*h,0]
```

X: $\int (g+h\,x)^m\,(a+b\,\text{Sin}[c+d\,(e+f\,x)^n])^p\,dx$

Rule:

$$\int (g+h\,x)^m\,(a+b\,\text{Sin}[c+d\,(e+f\,x)^n])^p\,dx \longrightarrow \int (g+h\,x)^m\,(a+b\,\text{Sin}[c+d\,(e+f\,x)^n])^p\,dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Sin[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  Unintegrable[(g+h*x)^m*(a+b*Sin[c+d*(e+f*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x]
```

```
Int[(g_.+h_.*x_)^m_.*(a_.+b_.*Cos[c_.+d_.*(e_.+f_.*x_)^n_])^p_.,x_Symbol] :=
  Unintegrable[(g+h*x)^m*(a+b*Cos[c+d*(e+f*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x]
```

N: $\int v^m \left(a+b\,\mathrm{Sin}\left[c+d\,u^n\right]\right)^p dx$ when $u == e+f\,x \land v == g+h\,x$

Derivation: Algebraic normalization

Rule: If $u == e+f\,x \land v == g+h\,x$, then

$$\int v^m \left(a+b\,\mathrm{Sin}\left[c+d\,u^n\right]\right)^p dx \to \int (g+h\,x)^m \left(a+b\,\mathrm{Sin}\left[c+d\,(e+f\,x)^n\right]\right)^p dx$$

Program code:

```
Int[v_^m_.*(a_.+b_.*Sin[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  Int[ExpandToSum[v,x]^m*(a+b*Sin[c+d*ExpandToSum[u,x]^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && LinearQ[u,x] && LinearQ[v,x] && Not[LinearMatchQ[u,x] && LinearMatchQ[v,x]]
```

```
Int[v_^m_.*(a_.+b_.*Cos[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  Int[ExpandToSum[v,x]^m*(a+b*Cos[c+d*ExpandToSum[u,x]^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && LinearQ[u,x] && LinearQ[v,x] && Not[LinearMatchQ[u,x] && LinearMatchQ[v,x]]
```

Rules for integrands of the form $x^m\,\mathrm{Sin}[a+b\,x^n]^p\,\mathrm{Cos}[a+b\,x^n]$

1. $\int x^m\,\mathrm{Sin}\left[a+b\,x^n\right]^p \mathrm{Cos}\left[a+b\,x^n\right] dx$ when $p \neq -1$

   1: $\int x^{n-1}\,\mathrm{Sin}\left[a+b\,x^n\right]^p \mathrm{Cos}\left[a+b\,x^n\right] dx$ when $p \neq -1$

Derivation: Power rule for integration

Rule: If $p \neq -1$, then

$$\int x^{n-1} \operatorname{Sin}[a+b\,x^n]^p \operatorname{Cos}[a+b\,x^n]\, dx \longrightarrow \frac{\operatorname{Sin}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)}$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_.*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
  Sin[a+b*x^n]^(p+1)/(b*n*(p+1)) /;
FreeQ[{a,b,m,n,p},x] && EqQ[m,n-1] && NeQ[p,-1]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_.*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
  -Cos[a+b*x^n]^(p+1)/(b*n*(p+1)) /;
FreeQ[{a,b,m,n,p},x] && EqQ[m,n-1] && NeQ[p,-1]
```

2: $\int x^m \operatorname{Sin}[a+b\,x^n]^p \operatorname{Cos}[a+b\,x^n]\,dx$ when $0<n<m+1\,\wedge\,p\neq -1$

Reference: G&R 2.645.6

Reference: G&R 2.645.3

Derivation: Integration by parts

Basis: $x^m \operatorname{Sin}[a+b\,x^n]^p \operatorname{Cos}[a+b\,x^n] == x^{m-n+1}\,\partial_x \frac{\operatorname{Sin}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)}$

Rule: If $0<n<m+1\,\wedge\,p\neq -1$, then

$$\int x^m \operatorname{Sin}[a+b\,x^n]^p \operatorname{Cos}[a+b\,x^n]\,dx \longrightarrow \frac{x^{m-n+1}\operatorname{Sin}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)}-\frac{m-n+1}{b\,n\,(p+1)}\int x^{m-n}\operatorname{Sin}[a+b\,x^n]^{p+1}\,dx$$

Program code:

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_.*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Sin[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*Sin[a+b*x^n]^(p+1),x] /;
FreeQ[{a,b,p},x] && LtQ[0,n,m+1] && NeQ[p,-1]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_.*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Cos[a+b*x^n]^(p+1)/(b*n*(p+1)) +
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*Cos[a+b*x^n]^(p+1),x] /;
FreeQ[{a,b,p},x] && LtQ[0,n,m+1] && NeQ[p,-1]
```