Rules for integrands involving inverse hyperbolic tangents and cotangents

1. $\int u \operatorname{ArcTanh}[a + b x^n] \, dx$

1: $\int \operatorname{ArcTanh}[a + b x^n] \, dx$

Derivation: Integration by parts

Rule:

$$\int \operatorname{ArcTanh}[a + b x^n] \, dx \longrightarrow x \operatorname{ArcTanh}[a + b x^n] - b n \int \frac{x^n}{1 - a^2 - 2 a b x^n - b^2 x^{2 n}} \, dx$$

Program code:

```
Int[ArcTanh[a_+b_.*x_^n_],x_Symbol] :=
  x*ArcTanh[a+b*x^n] -
  b*n*Int[x^n/(1-a^2-2*a*b*x^n-b^2*x^(2*n)),x] /;
FreeQ[{a,b,n},x]
```

```
Int[ArcCoth[a_+b_.*x_^n_],x_Symbol] :=
  x*ArcCoth[a+b*x^n] -
  b*n*Int[x^n/(1-a^2-2*a*b*x^n-b^2*x^(2*n)),x] /;
FreeQ[{a,b,n},x]
```

2. $\int x^m \text{ArcTanh}[a+b x^n] \, dx$

1: $\int \frac{\text{ArcTanh}[a+b x^n]}{x} \, dx$

Derivation: Algebraic expansion

Basis: $\text{ArcTanh}[z] == \frac{1}{2} \text{Log}[1+z] - \frac{1}{2} \text{Log}[1-z]$

Basis: $\text{ArcCoth}[z] == \frac{1}{2} \text{Log}\left[1+\frac{1}{z}\right] - \frac{1}{2} \text{Log}\left[1-\frac{1}{z}\right]$

Rule:

$$\int \frac{\text{ArcTanh}[a+b x^n]}{x} \, dx \longrightarrow \frac{1}{2} \int \frac{\text{Log}[1+a+b x^n]}{x} \, dx - \frac{1}{2} \int \frac{\text{Log}[1-a-b x^n]}{x} \, dx$$

Program code:

```
Int[ArcTanh[a_.+b_.*x_^n_.]/x_,x_Symbol] :=
  1/2*Int[Log[1+a+b*x^n]/x,x] -
  1/2*Int[Log[1-a-b*x^n]/x,x] /;
FreeQ[{a,b,n},x]
```

```
Int[ArcCoth[a_.+b_.*x_^n_.]/x_,x_Symbol] :=
  1/2*Int[Log[1+1/(a+b*x^n)]/x,x] -
  1/2*Int[Log[1-1/(a+b*x^n)]/x,x] /;
FreeQ[{a,b,n},x]
```

2: $\int x^m \operatorname{ArcTanh}[a + b x^n] \, dx$ when $(m \mid n) \in \mathbb{Q} \land m + 1 \neq 0 \land m + 1 \neq n$

Reference: CRC 588, A&S 4.6.54

Reference: CRC 590, A&S 4.6.60

Derivation: Integration by parts

Rule: If $(m \mid n) \in \mathbb{Q} \land m + 1 \neq 0 \land m + 1 \neq n$, then

$$\int x^m \operatorname{ArcTanh}[a + b x^n] \, dx \longrightarrow \frac{x^{m+1} \operatorname{ArcTanh}[a + b x^n]}{m + 1} - \frac{b n}{m + 1} \int \frac{x^{m+n}}{1 - a^2 - 2 a b x^n - b^2 x^{2 n}} \, dx$$

Program code:

```
Int[x_^m_.*ArcTanh[a_+b_.*x_^n_],x_Symbol] :=
  x^(m+1)*ArcTanh[a+b*x^n]/(m+1) -
  b*n/(m+1)*Int[x^(m+n)/(1-a^2-2*a*b*x^n-b^2*x^(2*n)),x] /;
FreeQ[{a,b},x] && RationalQ[m,n] && NeQ[m,-1] && NeQ[m+1,n]
```

```
Int[x_^m_.*ArcCoth[a_+b_.*x_^n_],x_Symbol] :=
  x^(m+1)*ArcCoth[a+b*x^n]/(m+1) -
  b*n/(m+1)*Int[x^(m+n)/(1-a^2-2*a*b*x^n-b^2*x^(2*n)),x] /;
FreeQ[{a,b},x] && RationalQ[m,n] && NeQ[m,-1] && NeQ[m+1,n]
```

2. $\int u \operatorname{ArcTanh}\left[a+b\, f^{c+d\,x}\right] dx$

1: $\int \operatorname{ArcTanh}\left[a+b\, f^{c+d\,x}\right] dx$

Derivation: Algebraic expansion

Basis: $\operatorname{ArcTanh}[z] = \frac{1}{2}\operatorname{Log}[1+z] - \frac{1}{2}\operatorname{Log}[1-z]$

Basis: $\operatorname{ArcCoth}[z] = \frac{1}{2}\operatorname{Log}\left[1+\frac{1}{z}\right] - \frac{1}{2}\operatorname{Log}\left[1-\frac{1}{z}\right]$

Rule:

$$\int \operatorname{ArcTanh}\left[a+b\, f^{c+d\,x}\right] dx \rightarrow \frac{1}{2}\int \operatorname{Log}\left[1+a+b\, f^{c+d\,x}\right] dx - \frac{1}{2}\int \operatorname{Log}\left[1-a-b\, f^{c+d\,x}\right] dx$$

Program code:

```
Int[ArcTanh[a_.+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  1/2*Int[Log[1+a+b*f^(c+d*x)],x] -
  1/2*Int[Log[1-a-b*f^(c+d*x)],x] /;
FreeQ[{a,b,c,d,f},x]
```

```
Int[ArcCoth[a_.+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  1/2*Int[Log[1+1/(a+b*f^(c+d*x))],x] -
  1/2*Int[Log[1-1/(a+b*f^(c+d*x))],x] /;
FreeQ[{a,b,c,d,f},x]
```

2: $\int x^m \operatorname{ArcTanh}\left[a+b\,f^{c+d\,x}\right] dx$ when $m \in \mathbb{Z} \wedge m > 0$

Derivation: Algebraic expansion

Basis: $\operatorname{ArcTanh}[z] == \frac{1}{2} \operatorname{Log}[1+z] - \frac{1}{2} \operatorname{Log}[1-z]$

Basis: $\operatorname{ArcCoth}[z] == \frac{1}{2} \operatorname{Log}\left[1+\frac{1}{z}\right] - \frac{1}{2} \operatorname{Log}\left[1-\frac{1}{z}\right]$

Rule: If $m \in \mathbb{Z} \wedge m > 0$, then

$$\int x^m \operatorname{ArcTanh}\left[a+b\,f^{c+d\,x}\right] dx \longrightarrow \frac{1}{2} \int x^m \operatorname{Log}\left[1+a+b\,f^{c+d\,x}\right] dx - \frac{1}{2} \int x^m \operatorname{Log}\left[1-a-b\,f^{c+d\,x}\right] dx$$

Program code:

```
Int[x_^m_.*ArcTanh[a_.+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  1/2*Int[x^m*Log[1+a+b*f^(c+d*x)],x] -
  1/2*Int[x^m*Log[1-a-b*f^(c+d*x)],x] /;
FreeQ[{a,b,c,d,f},x] && IGtQ[m,0]
```

```
Int[x_^m_.*ArcCoth[a_.+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  1/2*Int[x^m*Log[1+1/(a+b*f^(c+d*x))],x] -
  1/2*Int[x^m*Log[1-1/(a+b*f^(c+d*x))],x] /;
FreeQ[{a,b,c,d,f},x] && IGtQ[m,0]
```

3: $\int u\,\text{ArcTanh}\left[\frac{c}{a+b\,x^n}\right]^m dx$

Derivation: Algebraic simplification

Basis: $\text{ArcTanh}[z] == \text{ArcCoth}\left[\frac{1}{z}\right]$

Rule:

$$\int u\,\text{ArcTanh}\left[\frac{c}{a+b\,x^n}\right]^m dx \to \int u\,\text{ArcCoth}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

Program code:

```
Int[u_.*ArcTanh[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCoth[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

```
Int[u_.*ArcCoth[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcTanh[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

4. $\int u\,\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] dx$ when $b == c^2$

1: $\int \text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] dx$ when $b == c^2$

Derivation: Integration by parts

Basis: If $b == c^2$, then $\partial_x \text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] == \frac{c}{\sqrt{a+b\,x^2}}$

Rule: If $b == c^2$, then

$$\int \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] dx \longrightarrow x\operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] - c\int \frac{x}{\sqrt{a+b\,x^2}}\,dx$$

Program code:

```
Int[ArcTanh[c_.*x_/Sqrt[a_.+b_.*x_^2]],x_Symbol] :=
  x*ArcTanh[(c*x)/Sqrt[a+b*x^2]] - c*Int[x/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c},x] && EqQ[b,c^2]
```

```
Int[ArcCoth[c_.*x_/Sqrt[a_.+b_.*x_^2]],x_Symbol] :=
  x*ArcCoth[(c*x)/Sqrt[a+b*x^2]] - c*Int[x/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c},x] && EqQ[b,c^2]
```

2. $\int (d\,x)^m \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] dx$ when $b == c^2$

1: $\int \frac{\operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]}{x}\,dx$ when $b == c^2$

Derivation: Integration by parts

Basis: If $b == c^2$, then $\partial_x \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] == \frac{c}{\sqrt{a+b\,x^2}}$

Rule: If $b == c^2$, then

$$\int \frac{\operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]}{x}\,dx \longrightarrow \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]\operatorname{Log}[x] - c\int \frac{\operatorname{Log}[x]}{\sqrt{a+b\,x^2}}\,dx$$

Program code:

```
Int[ArcTanh[c_.*x_/Sqrt[a_.+b_.*x_^2]]/x_,x_Symbol] :=
  ArcTanh[c*x/Sqrt[a+b*x^2]]*Log[x] - c*Int[Log[x]/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c},x] && EqQ[b,c^2]
```

```
Int[ArcCoth[c_.*x_/Sqrt[a_.+b_.*x_^2]]/x_,x_Symbol] :=
  ArcCoth[c*x/Sqrt[a+b*x^2]]*Log[x] - c*Int[Log[x]/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c},x] && EqQ[b,c^2]
```

2: $\int (d\,x)^m \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] dx$ when $b == c^2 \land m \neq -1$

Derivation: Integration by parts

Basis: If $b == c^2$, then $\partial_x \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] == \frac{c}{\sqrt{a+b\,x^2}}$

Rule: If $b == c^2 \land m \neq -1$, then

$$\int (d\,x)^m \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] dx \to \frac{(d\,x)^{m+1} \operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]}{d\,(m+1)} - \frac{c}{d\,(m+1)} \int \frac{(d\,x)^{m+1}}{\sqrt{a+b\,x^2}}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*ArcTanh[c_.*x_/Sqrt[a_.+b_.*x_^2]],x_Symbol] :=
  (d*x)^(m+1)*ArcTanh[(c*x)/Sqrt[a+b*x^2]]/(d*(m+1)) - c/(d*(m+1))*Int[(d*x)^(m+1)/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b,c^2] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*ArcCoth[c_.*x_/Sqrt[a_.+b_.*x_^2]],x_Symbol] :=
  (d*x)^(m+1)*ArcCoth[(c*x)/Sqrt[a+b*x^2]]/(d*(m+1)) - c/(d*(m+1))*Int[(d*x)^(m+1)/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b,c^2] && NeQ[m,-1]
```

3. $\int \frac{\operatorname{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^m}{\sqrt{d+e\,x^2}}\,dx$ when $b == c^2 \land b\,d - a\,e == 0$

1. $\int \frac{\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^m}{\sqrt{a+b\,x^2}}\,dx$ when $b == c^2$

1: $\int \frac{1}{\sqrt{a+b\,x^2}\;\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]}\,dx$ when $b == c^2$

Derivation: Reciprocal rule for integration

Basis: If $b == c^2$, then $\partial_x \text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] == \frac{c}{\sqrt{a+b\,x^2}}$

Rule: If $b == c^2$, then

$$\int \frac{1}{\sqrt{a+b\,x^2}\;\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]}\,dx \to \frac{1}{c}\,\text{Log}\left[\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]\right]$$

Program code:

```
Int[1/(Sqrt[a_.+b_.*x_^2]*ArcTanh[c_.*x_/Sqrt[a_.+b_.*x_^2]]),x_Symbol] :=
  1/c*Log[ArcTanh[c*x/Sqrt[a+b*x^2]]] /;
FreeQ[{a,b,c},x] && EqQ[b,c^2]
```

```
Int[1/(Sqrt[a_.+b_.*x_^2]*ArcCoth[c_.*x_/Sqrt[a_.+b_.*x_^2]]),x_Symbol] :=
  -1/c*Log[ArcCoth[c*x/Sqrt[a+b*x^2]]] /;
FreeQ[{a,b,c},x] && EqQ[b,c^2]
```

2: $\int \frac{\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^m}{\sqrt{a+b\,x^2}}\,dx$ when $b == c^2 \land m \neq -1$

Derivation: Power rule for integration

Basis: If $b == c^2$, then $\partial_x \text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right] == \frac{c}{\sqrt{a+b\,x^2}}$

Rule: If $b == c^2 \wedge m \neq -1$, then

$$\int \frac{\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^m}{\sqrt{a+b\,x^2}}\,dx \longrightarrow \frac{\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^{m+1}}{c\,(m+1)}$$

Program code:

```
Int[ArcTanh[c_.*x_/Sqrt[a_.+b_.*x_^2]]^m_./Sqrt[a_.+b_.*x_^2],x_Symbol] :=
  ArcTanh[c*x/Sqrt[a+b*x^2]]^(m+1)/(c*(m+1)) /;
FreeQ[{a,b,c,m},x] && EqQ[b,c^2] && NeQ[m,-1]
```

```
Int[ArcCoth[c_.*x_/Sqrt[a_.+b_.*x_^2]]^m_./Sqrt[a_.+b_.*x_^2],x_Symbol] :=
  -ArcCoth[c*x/Sqrt[a+b*x^2]]^(m+1)/(c*(m+1)) /;
FreeQ[{a,b,c,m},x] && EqQ[b,c^2] && NeQ[m,-1]
```

2: $\int \frac{\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^m}{\sqrt{d+e\,x^2}}\,dx$ when $b == c^2 \land b\,d-a\,e == 0$

Derivation: Piecewise constant extraction

Basis: If $b\,d-a\,e == 0$, then $\partial_x \frac{\sqrt{a+b\,x^2}}{\sqrt{d+e\,x^2}} == 0$

Rule: If $b == c^2 \land b\,d-a\,e == 0$, then

$$\int \frac{\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^m}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{\sqrt{a+b\,x^2}}{\sqrt{d+e\,x^2}} \int \frac{\text{ArcTanh}\left[\frac{c\,x}{\sqrt{a+b\,x^2}}\right]^m}{\sqrt{a+b\,x^2}}\,dx$$

Program code:

```
Int[ArcTanh[c_.*x_/Sqrt[a_.+b_.*x_^2]]^m_./Sqrt[d_.+e_.*x_^2],x_Symbol] :=
  Sqrt[a+b*x^2]/Sqrt[d+e*x^2]*Int[ArcTanh[c*x/Sqrt[a+b*x^2]]^m/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[b,c^2] && EqQ[b*d-a*e,0]
```

```
Int[ArcCoth[c_.*x_/Sqrt[a_.+b_.*x_^2]]^m_./Sqrt[d_.+e_.*x_^2],x_Symbol] :=
  Sqrt[a+b*x^2]/Sqrt[d+e*x^2]*Int[ArcCoth[c*x/Sqrt[a+b*x^2]]^m/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[b,c^2] && EqQ[b*d-a*e,0]
```

5: $\int \frac{f[x, \text{ArcTanh}[a+b\,x]]}{1-(a+b\,x)^2}\,dx$

Derivation: Integration by substitution

Basis: $\frac{f[z]}{1-z^2} == f[\text{Tanh}[\text{ArcTanh}[z]]]\ \text{ArcTanh}'[z]$

Basis: $r+s\,x+t\,x^2 == -\frac{s^2-4\,r\,t}{4\,t}\left(1-\frac{(s+2\,t\,x)^2}{s^2-4\,r\,t}\right)$

Basis: $1-\text{Tanh}[z]^2 == \text{Sech}[z]^2$

Rule:

$$\int \frac{f[x, \text{ArcTanh}[a+b\,x]]}{1-(a+b\,x)^2}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int f\left[-\frac{a}{b}+\frac{\text{Tanh}[x]}{b}, x\right]dx, x, \text{ArcTanh}[a+b\,x]\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_*v_^n_.,x_Symbol] :=
  With[{tmp=InverseFunctionOfLinear[u,x]},
  ShowStep["","Int[f[x,ArcTanh[a+b*x]]/(1-(a+b*x)^2),x]",
           "Subst[Int[f[-a/b+Tanh[x]/b,x],x],x,ArcTanh[a+b*x]]/b",Hold[
  (-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1]*
    Subst[Int[SimplifyIntegrand[SubstForInverseFunction[u,tmp,x]*Sech[x]^(2*(n+1)),x],x], x, tmp]]] /;
 Not[FalseQ[tmp]] && EqQ[Head[tmp],ArcTanh] && EqQ[Discriminant[v,x]*tmp[[1]]^2-D[v,x]^2,0]] /;
SimplifyFlag && QuadraticQ[v,x] && ILtQ[n,0] && PosQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^w_ /; FreeQ[f,x]],

Int[u_*v_^n_.,x_Symbol] :=
  With[{tmp=InverseFunctionOfLinear[u,x]},
  (-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1]*
    Subst[Int[SimplifyIntegrand[SubstForInverseFunction[u,tmp,x]*Sech[x]^(2*(n+1)),x],x], x, tmp] /;
 Not[FalseQ[tmp]] && EqQ[Head[tmp],ArcTanh] && EqQ[Discriminant[v,x]*tmp[[1]]^2-D[v,x]^2,0]] /;
QuadraticQ[v,x] && ILtQ[n,0] && PosQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^w_ /; FreeQ[f,x]]]
```

```
If[TrueQ[$LoadShowSteps],

Int[u_*v_^n_.,x_Symbol] :=
  With[{tmp=InverseFunctionOfLinear[u,x]},
  ShowStep["","Int[f[x,ArcCoth[a+b*x]]/(1-(a+b*x)^2),x]",
           "Subst[Int[f[-a/b+Coth[x]/b,x],x],x,ArcCoth[a+b*x]]/b",Hold[
  (-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1]*
    Subst[Int[SimplifyIntegrand[SubstForInverseFunction[u,tmp,x]*(-Csch[x]^2)^(n+1),x],x], x, tmp]]] /;
 Not[FalseQ[tmp]] && EqQ[Head[tmp],ArcCoth] && EqQ[Discriminant[v,x]*tmp[[1]]^2-D[v,x]^2,0]] /;
SimplifyFlag && QuadraticQ[v,x] && ILtQ[n,0] && PosQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^w_ /; FreeQ[f,x]],

Int[u_*v_^n_.,x_Symbol] :=
  With[{tmp=InverseFunctionOfLinear[u,x]},
  (-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1]*
    Subst[Int[SimplifyIntegrand[SubstForInverseFunction[u,tmp,x]*(-Csch[x]^2)^(n+1),x],x], x, tmp] /;
 Not[FalseQ[tmp]] && EqQ[Head[tmp],ArcCoth] && EqQ[Discriminant[v,x]*tmp[[1]]^2-D[v,x]^2,0]] /;
QuadraticQ[v,x] && ILtQ[n,0] && PosQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^w_ /; FreeQ[f,x]]]
```

6. $\int u \operatorname{ArcTanh}[c + d \operatorname{Tanh}[a + b x]] \, dx$

1. $\int \operatorname{ArcTanh}[c + d \operatorname{Tanh}[a + b x]] \, dx$

1: $\int \operatorname{ArcTanh}[c + d \operatorname{Tanh}[a + b x]] \, dx$ when $(c - d)^2 == 1$

Derivation: Integration by parts

Basis: If $(c - d)^2 == 1$, then $\partial_x \operatorname{ArcTanh}[c + d \operatorname{Tanh}[a + b x]] == -\frac{b}{c-d+c e^{2 a+2 b x}}$

Rule: If $(c - d)^2 == 1$, then

$$\int \operatorname{ArcTanh}[c + d \operatorname{Tanh}[a + b x]] \, dx \longrightarrow x \operatorname{ArcTanh}[c + d \operatorname{Tanh}[a + b x]] + b \int \frac{x}{c - d + c e^{2 a+2 b x}} \, dx$$

Program code:

```
Int[ArcTanh[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Tanh[a+b*x]] +
  b*Int[x/(c-d+c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c-d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Tanh[a+b*x]] +
  b*Int[x/(c-d+c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c-d)^2,1]
```

```
Int[ArcTanh[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Coth[a+b*x]] +
  b*Int[x/(c-d-c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c-d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Coth[a+b*x]] +
  b*Int[x/(c-d-c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c-d)^2,1]
```

2: $\int \text{ArcTanh}[c+d\,\text{Tanh}[a+b\,x]]\,dx$ when $(c-d)^2 \neq 1$

Derivation: Integration by parts

Basis: $\partial_x \text{ArcTanh}[c+d\,\text{Tanh}[a+b\,x]] == -\frac{b\,(1-c-d)\,e^{2a+2bx}}{1-c+d+(1-c-d)\,e^{2\,(a+b\,x)}} + \frac{b\,(1+c+d)\,e^{2a+2bx}}{1+c-d+(1+c+d)\,e^{2a+2bx}}$

Rule: If $(c-d)^2 \neq 1$, then

$$\int \text{ArcTanh}[c+d\,\text{Tanh}[a+b\,x]]\,dx \longrightarrow$$
$$x\,\text{ArcTanh}[c+d\,\text{Tanh}[a+b\,x]] + b\,(1-c-d)\int \frac{x\,e^{2a+2bx}}{1-c+d+(1-c-d)\,e^{2a+2bx}}\,dx - b\,(1+c+d)\int \frac{x\,e^{2a+2bx}}{1+c-d+(1+c+d)\,e^{2a+2bx}}\,dx$$

Basis: $\partial_x \text{ArcTanh}[c+d\,\text{Tanh}[a+b\,x]] == -\frac{b\,(1+c-d)}{1+c-d+(1+c+d)\,e^{2a+2bx}} + \frac{b\,(1-c+d)}{1-c+d+(1-c-d)\,e^{2a+2bx}}$

Note: Although this formula appears simpler, it either introduces superfluous terms that have to be cancelled out, or results in a slightly more complicated antiderivative.

Rule: If $(c-d)^2 \neq 1$, then

$$\int \text{ArcTanh}[c+d\,\text{Tanh}[a+b\,x]]\,dx \longrightarrow$$
$$x\,\text{ArcTanh}[c+d\,\text{Tanh}[a+b\,x]] + b\,(1+c-d)\int \frac{x}{1+c-d+(1+c+d)\,e^{2a+2bx}}\,dx - b\,(1-c+d)\int \frac{x}{1-c+d+(1-c-d)\,e^{2a+2bx}}\,dx$$

Program code:

```
Int[ArcTanh[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Tanh[a+b*x]] +
  b*(1-c-d)*Int[x*E^(2*a+2*b*x)/(1-c+d+(1-c-d)*E^(2*a+2*b*x)),x] -
  b*(1+c+d)*Int[x*E^(2*a+2*b*x)/(1+c-d+(1+c+d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c-d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Tanh[a+b*x]] +
  b*(1-c-d)*Int[x*E^(2*a+2*b*x)/(1-c+d+(1-c-d)*E^(2*a+2*b*x)),x] -
  b*(1+c+d)*Int[x*E^(2*a+2*b*x)/(1+c-d+(1+c+d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c-d)^2,1]
```

```
Int[ArcTanh[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Coth[a+b*x]] +
  b*(1+c+d)*Int[x*E^(2*a+2*b*x)/(1+c-d-(1+c+d)*E^(2*a+2*b*x)),x] -
  b*(1-c-d)*Int[x*E^(2*a+2*b*x)/(1-c+d-(1-c-d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c-d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Coth[a+b*x]] +
  b*(1+c+d)*Int[x*E^(2*a+2*b*x)/(1+c-d-(1+c+d)*E^(2*a+2*b*x)),x] -
  b*(1-c-d)*Int[x*E^(2*a+2*b*x)/(1-c+d-(1-c-d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c-d)^2,1]
```

2. $\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]] \, dx$ when $m \in \mathbb{Z}^+$

1: $\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]] \, dx$ when $m \in \mathbb{Z}^+ \land (c-d)^2 == 1$

Derivation: Integration by parts

Basis: If $(c-d)^2 == 1$, then $\partial_x \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]] == -\frac{b}{c-d+c e^{2 a+2 b x}}$

Rule: If $m \in \mathbb{Z}^+ \land (c-d)^2 == 1$, then

$$\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]] \, dx \to \frac{(e+f x)^{m+1} \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]]}{f (m+1)} + \frac{b}{f (m+1)} \int \frac{(e+f x)^{m+1}}{c-d+c e^{2 a+2 b x}} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Tanh[a+b*x]]/(f*(m+1)) +
  b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c-d+c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c-d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Tanh[a+b*x]]/(f*(m+1)) +
  b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c-d+c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c-d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Coth[a+b*x]]/(f*(m+1)) +
  b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c-d-c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c-d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Coth[a+b*x]]/(f*(m+1)) +
  b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c-d-c*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c-d)^2,1]
```

2: $\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]] \, dx$ when $m \in \mathbb{Z}^+ \wedge (c-d)^2 \neq 1$

Derivation: Integration by parts

Basis: $\partial_x \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]] = -\frac{b (1-c-d) e^{2 a+2 b x}}{1-c+d+(1-c-d) e^{2 (a+b x)}} + \frac{b (1+c+d) e^{2 a+2 b x}}{1+c-d+(1+c+d) e^{2 a+2 b x}}$

Rule: If $m \in \mathbb{Z}^+ \wedge (c-d)^2 \neq 1$, then

$$\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]] \, dx \to$$
$$\frac{(e+f x)^{m+1} \operatorname{ArcTanh}[c+d \operatorname{Tanh}[a+b x]]}{f (m+1)} + \frac{b (1-c-d)}{f (m+1)} \int \frac{(e+f x)^{m+1} e^{2 a+2 b x}}{1-c+d+(1-c-d) e^{2 a+2 b x}} \, dx - \frac{b (1+c+d)}{f (m+1)} \int \frac{(e+f x)^{m+1} e^{2 a+2 b x}}{1+c-d+(1+c+d) e^{2 a+2 b x}} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Tanh[a+b*x]]/(f*(m+1)) +
  b*(1-c-d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1-c+d+(1-c-d)*E^(2*a+2*b*x)),x] -
  b*(1+c+d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1+c-d+(1+c+d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c-d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Tanh[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Tanh[a+b*x]]/(f*(m+1)) +
  b*(1-c-d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1-c+d+(1-c-d)*E^(2*a+2*b*x)),x] -
  b*(1+c+d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1+c-d+(1+c+d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c-d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Coth[a+b*x]]/(f*(m+1)) +
  b*(1+c+d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1+c-d-(1+c+d)*E^(2*a+2*b*x)),x] -
  b*(1-c-d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1-c+d-(1-c-d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c-d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Coth[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Coth[a+b*x]]/(f*(m+1)) +
  b*(1+c+d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1+c-d-(1+c+d)*E^(2*a+2*b*x)),x] -
  b*(1-c-d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*a+2*b*x)/(1-c+d-(1-c-d)*E^(2*a+2*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c-d)^2,1]
```

7. $\int u\,\text{ArcTanh}[c+d\,\text{Tan}[a+b\,x]]\,dx$

1. $\int u\,\text{ArcTanh}[\text{Tan}[a+b\,x]]\,dx$

1: $\int \text{ArcTanh}[\text{Tan}[a+b\,x]]\,dx$

Derivation: Integration by parts

Basis: $\partial_x \text{ArcTanh}[\text{Tan}[a+b\,x]] == b\,\text{Sec}[2\,a+2\,b\,x]$

Rule:

$$\int \text{ArcTanh}[\text{Tan}[a+b\,x]]\,dx \longrightarrow x\,\text{ArcTanh}[\text{Tan}[a+b\,x]] - b\int x\,\text{Sec}[2\,a+2\,b\,x]\,dx$$

Program code:

```
Int[ArcTanh[Tan[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[Tan[a+b*x]] - b*Int[x*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b},x]
```

```
Int[ArcCoth[Tan[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[Tan[a+b*x]] - b*Int[x*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b},x]
```

```
Int[ArcTanh[Cot[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[Cot[a+b*x]] - b*Int[x*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b},x]
```

```
Int[ArcCoth[Cot[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[Cot[a+b*x]] - b*Int[x*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b},x]
```

2: $\int (e+f x)^m \operatorname{ArcTanh}[\operatorname{Tan}[a+b x]]\, dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x \operatorname{ArcTanh}[\operatorname{Tan}[a+b x]] = b \operatorname{Sec}[2 a+2 b x]$

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (e+f x)^m \operatorname{ArcTanh}[\operatorname{Tan}[a+b x]]\, dx \to \frac{(e+f x)^{m+1} \operatorname{ArcTanh}[\operatorname{Tan}[a+b x]]}{f (m+1)} - \frac{b}{f (m+1)} \int (e+f x)^{m+1} \operatorname{Sec}[2 a+2 b x]\, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[Tan[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[Tan[a+b*x]]/(f*(m+1)) - b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b,e,f},x] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[Tan[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[Tan[a+b*x]]/(f*(m+1)) - b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b,e,f},x] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[Cot[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[Cot[a+b*x]]/(f*(m+1)) - b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b,e,f},x] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[Cot[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[Cot[a+b*x]]/(f*(m+1)) - b/(f*(m+1))*Int[(e+f*x)^(m+1)*Sec[2*a+2*b*x],x] /;
FreeQ[{a,b,e,f},x] && IGtQ[m,0]
```

2. $\int u\,\text{ArcTanh}[c+d\,\text{Tan}[a+b\,x]]\,dx$

   1. $\int \text{ArcTanh}[c+d\,\text{Tan}[a+b\,x]]\,dx$

      1: $\int \text{ArcTanh}[c+d\,\text{Tan}[a+b\,x]]\,dx$ when $(c+i\,d)^2 == 1$

Derivation: Integration by parts

Basis: If $(c+i\,d)^2 == 1$, then $\partial_x \text{ArcTanh}[c+d\,\text{Tan}[a+b\,x]] == -\frac{i\,b}{c+i\,d+c\,e^{2\,i\,a+2\,i\,b\,x}}$

Rule: If $(c+i\,d)^2 == 1$, then

$$\int \text{ArcTanh}[c+d\,\text{Tan}[a+b\,x]]\,dx \longrightarrow x\,\text{ArcTanh}[c+d\,\text{Tan}[a+b\,x]]+i\,b\int \frac{x}{c+i\,d+c\,e^{2\,i\,a+2\,i\,b\,x}}\,dx$$

Program code:

```
Int[ArcTanh[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Tan[a+b*x]] +
  I*b*Int[x/(c+I*d+c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c+I*d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Tan[a+b*x]] +
  I*b*Int[x/(c+I*d+c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c+I*d)^2,1]
```

```
Int[ArcTanh[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Cot[a+b*x]] +
  I*b*Int[x/(c-I*d-c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c-I*d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Cot[a+b*x]] +
  I*b*Int[x/(c-I*d-c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[(c-I*d)^2,1]
```

2: $\int \text{ArcTanh}[c + d \,\text{Tan}[a + b\, x]]\, dx$ when $(c + i\, d)^2 \neq 1$

Derivation: Integration by parts

Basis: $\partial_x \text{ArcTanh}[c + d \,\text{Tan}[a + b\, x]] == -\frac{i\, b\, (1-c+i\, d)\, e^{2 i a+2 i b x}}{1-c-i\, d+(1-c+i\, d)\, e^{2 i a+2 i b x}} + \frac{i\, b\, (1+c-i\, d)\, e^{2 i a+2 i b x}}{1+c+i\, d+(1+c-i\, d)\, e^{2 i a+2 i b x}}$

Rule: If $(c + i\, d)^2 \neq 1$, then

$$\int \text{ArcTanh}[c + d\, \text{Tan}[a + b\, x]]\, dx \longrightarrow$$

$$x\, \text{ArcTanh}[c + d\, \text{Tan}[a + b\, x]] + i\, b\, (1 - c + i\, d) \int \frac{x\, e^{2 i a+2 i b x}}{1 - c - i\, d + (1 - c + i\, d)\, e^{2 i a+2 i b x}}\, dx - i\, b\, (1 + c - i\, d) \int \frac{x\, e^{2 i a+2 i b x}}{1 + c + i\, d + (1 + c - i\, d)\, e^{2 i a+2 i b x}}\, dx$$

Program code:

```
Int[ArcTanh[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Tan[a+b*x]] +
  I*b*(1-c+I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1-c-I*d+(1-c+I*d)*E^(2*I*a+2*I*b*x)),x] -
  I*b*(1+c-I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1+c+I*d+(1+c-I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c+I*d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Tan[a+b*x]] +
  I*b*(1-c+I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1-c-I*d+(1-c+I*d)*E^(2*I*a+2*I*b*x)),x] -
  I*b*(1+c-I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1+c+I*d+(1+c-I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c+I*d)^2,1]
```

```
Int[ArcTanh[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  x*ArcTanh[c+d*Cot[a+b*x]] -
  I*b*(1-c-I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1-c+I*d-(1-c-I*d)*E^(2*I*a+2*I*b*x)),x] +
  I*b*(1+c+I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1+c-I*d-(1+c+I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c-I*d)^2,1]
```

```
Int[ArcCoth[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  x*ArcCoth[c+d*Cot[a+b*x]] -
  I*b*(1-c-I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1-c+I*d-(1-c-I*d)*E^(2*I*a+2*I*b*x)),x] +
  I*b*(1+c+I*d)*Int[x*E^(2*I*a+2*I*b*x)/(1+c-I*d-(1+c+I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[(c-I*d)^2,1]
```

2. $\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]] \, dx$ when $m \in \mathbb{Z}^+$

   1: $\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]] \, dx$ when $m \in \mathbb{Z}^+ \land (c+i d)^2 == 1$

Derivation: Integration by parts

Basis: If $(c+i d)^2 == 1$, then $\partial_x \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]] == -\frac{i b}{c+i d+c e^{2 i a+2 i b x}}$

Rule: If $m \in \mathbb{Z}^+ \land (c+i d)^2 == 1$, then

$$\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]] \, dx \longrightarrow \frac{(e+f x)^{m+1} \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]]}{f (m+1)} + \frac{i b}{f (m+1)} \int \frac{(e+f x)^{m+1}}{c+i d+c e^{2 i a+2 i b x}} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Tan[a+b*x]]/(f*(m+1)) +
  I*b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c+I*d+c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c+I*d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Tan[a+b*x]]/(f*(m+1)) +
  I*b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c+I*d+c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c+I*d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Cot[a+b*x]]/(f*(m+1)) +
  I*b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c-I*d-c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c-I*d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Cot[a+b*x]]/(f*(m+1)) +
  I*b/(f*(m+1))*Int[(e+f*x)^(m+1)/(c-I*d-c*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[(c-I*d)^2,1]
```

2: $\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]] \, dx$ when $m \in \mathbb{Z}^+ \land (c+i d)^2 \neq 1$

Derivation: Integration by parts

Basis: $\partial_x \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]] == -\frac{i b (1-c+i d) e^{2 i a+2 i b x}}{1-c-i d+(1-c+i d) e^{2 i a+2 i b x}}+\frac{i b (1+c-i d) e^{2 i a+2 i b x}}{1+c+i d+(1+c-i d) e^{2 i a+2 i b x}}$

Rule: If $m \in \mathbb{Z}^+ \land (c+i d)^2 \neq 1$, then

$$\int (e+f x)^m \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]] \, dx \to$$

$$\frac{(e+f x)^{m+1} \operatorname{ArcTanh}[c+d \operatorname{Tan}[a+b x]]}{f (m+1)}+\frac{i b (1-c+i d)}{f (m+1)} \int \frac{(e+f x)^{m+1} e^{2 i a+2 i b x}}{1-c-i d+(1-c+i d) e^{2 i a+2 i b x}} \, dx-\frac{i b (1+c-i d)}{f (m+1)} \int \frac{(e+f x)^{m+1} e^{2 i a+2 i b x}}{1+c+i d+(1+c-i d) e^{2 i a+2 i b x}} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Tan[a+b*x]]/(f*(m+1)) +
  I*b*(1-c+I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1-c-I*d+(1-c+I*d)*E^(2*I*a+2*I*b*x)),x] -
  I*b*(1+c-I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1+c+I*d+(1+c-I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c+I*d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Tan[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Tan[a+b*x]]/(f*(m+1)) +
  I*b*(1-c+I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1-c-I*d+(1-c+I*d)*E^(2*I*a+2*I*b*x)),x] -
  I*b*(1+c-I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1+c+I*d+(1+c-I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c+I*d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcTanh[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcTanh[c+d*Cot[a+b*x]]/(f*(m+1)) -
  I*b*(1-c-I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1-c+I*d-(1-c-I*d)*E^(2*I*a+2*I*b*x)),x] +
  I*b*(1+c+I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1+c-I*d-(1+c+I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c-I*d)^2,1]
```

```
Int[(e_.+f_.*x_)^m_.*ArcCoth[c_.+d_.*Cot[a_.+b_.*x_]],x_Symbol] :=
  (e+f*x)^(m+1)*ArcCoth[c+d*Cot[a+b*x]]/(f*(m+1)) -
  I*b*(1-c-I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1-c+I*d-(1-c-I*d)*E^(2*I*a+2*I*b*x)),x] +
  I*b*(1+c+I*d)/(f*(m+1))*Int[(e+f*x)^(m+1)*E^(2*I*a+2*I*b*x)/(1+c-I*d-(1+c+I*d)*E^(2*I*a+2*I*b*x)),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[(c-I*d)^2,1]
```

8. $\int v\,(a + b\,\text{ArcTanh}[u])\,dx$ when **u is free of inverse functions**

1: $\int \text{ArcTanh}[u]\,dx$ when **u is free of inverse functions**

Derivation: Integration by parts

Rule: If u is free of inverse functions, then

$$\int \text{ArcTanh}[u]\,dx \longrightarrow x\,\text{ArcTanh}[u] - \int \frac{x\,\partial_x u}{1 - u^2}\,dx$$

Program code:

```
Int[ArcTanh[u_],x_Symbol] :=
  x*ArcTanh[u] -
  Int[SimplifyIntegrand[x*D[u,x]/(1-u^2),x],x] /;
InverseFunctionFreeQ[u,x]
```

```
Int[ArcCoth[u_],x_Symbol] :=
  x*ArcCoth[u] -
  Int[SimplifyIntegrand[x*D[u,x]/(1-u^2),x],x] /;
InverseFunctionFreeQ[u,x]
```

2: $\int (c + d\,x)^m\,(a + b\,\text{ArcTanh}[u])\,dx$ when $m \neq -1\,\wedge\,u$ is free of inverse functions

Derivation: Integration by parts

Rule: If $m \neq -1\,\wedge$ u is free of inverse functions, then

$$\int (c + d\,x)^m\,(a + b\,\text{ArcTanh}[u])\,dx \;\longrightarrow\; \frac{(c + d\,x)^{m+1}\,(a + b\,\text{ArcTanh}[u])}{d\,(m + 1)} - \frac{b}{d\,(m + 1)}\int \frac{(c + d\,x)^{m+1}\,\partial_x u}{1 - u^2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcTanh[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcTanh[u])/(d*(m+1)) -
  b/(d*(m+1))*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/(1-u^2),x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && FalseQ[PowerVariableExpn[u,m+
```

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcCoth[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcCoth[u])/(d*(m+1)) -
  b/(d*(m+1))*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/(1-u^2),x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && FalseQ[PowerVariableExpn[u,m+
```

3: $\int v\,(a+b\,\text{ArcTanh}[u])\,dx$ when **u** and $\int v\,dx$ **are free of inverse functions**

Derivation: Integration by parts

Rule: If u is free of inverse functions, let $w = \int v\,dx$, if w is free of inverse functions, then

$$\int v\,(a+b\,\text{ArcTanh}[u])\,dx \longrightarrow w\,(a+b\,\text{ArcTanh}[u]) - b\int\frac{w\,\partial_x u}{1-u^2}\,dx$$

Program code:

```
Int[v_*(a_.+b_.*ArcTanh[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcTanh[u]),w,x] - b*Int[SimplifyIntegrand[w*D[u,x]/(1-u^2),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]] && FalseQ[FunctionOfLinear[v*(a+b*A
```

```
Int[v_*(a_.+b_.*ArcCoth[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcCoth[u]),w,x] - b*Int[SimplifyIntegrand[w*D[u,x]/(1-u^2),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]] && FalseQ[FunctionOfLinear[v*(a+b*A
```