Rules for integrands of the form $(a+b x+c x^2)^p (d+e x+f x^2)^q$

1. $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $c d-a f=0 \land b d-a e=0$

1: $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $c d-a f=0 \land b d-a e=0 \land \left(p\in\mathbb{Z} \lor \frac{c}{f}>0\right)$

Derivation: Algebraic simplification

Basis: If $c d-a f=0 \land b d-a e=0 \land \left(p\in\mathbb{Z} \lor \frac{c}{f}>0\right)$, then $(a+b x+c x^2)^p = \left(\frac{c}{f}\right)^p (d+e x+f x^2)^p$

Rule 1.2.1.5.1.1: If $c d-a f=0 \land b d-a e=0 \land \left(p\in\mathbb{Z} \lor \frac{c}{f}>0\right)$, then

$$\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx \longrightarrow \left(\frac{c}{f}\right)^p \int (d+e x+f x^2)^{p+q} \, dx$$

Program code:

```
Int[(a_+b_.*x_+c_.*x_^2)^p_.*(d_+e_.*x_+f_.*x_^2)^q_.,x_Symbol] :=
  (c/f)^p*Int[(d+e*x+f*x^2)^(p+q),x] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && EqQ[c*d-a*f,0] && EqQ[b*d-a*e,0] && (IntegerQ[p] || GtQ[c/f,0]) &&
  (Not[IntegerQ[q]] || LeafCount[d+e*x+f*x^2]≤LeafCount[a+b*x+c*x^2])
```

2: $\int (a+b\,x+c\,x^2)^p\,(d+e\,x+f\,x^2)^q\,dx$ when $c\,d-a\,f==0\,\wedge\,b\,d-a\,e==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,q\notin\mathbb{Z}\,\wedge\,\frac{c}{f}\ngtr 0$

Derivation: Piecewise constant extraction

Basis: If $c\,d-a\,f==0\,\wedge\,b\,d-a\,e==0$, then $\partial_x\frac{(a+b\,x+c\,x^2)^p}{(d+e\,x+f\,x^2)^p}==0$

Basis: If $c\,d-a\,f==0\,\wedge\,b\,d-a\,e==0$, then $\frac{(a+b\,x+c\,x^2)^p}{(d+e\,x+f\,x^2)^p}==\frac{a^{\text{IntPart}[p]}\,(a+b\,x+c\,x^2)^{\text{FracPart}[p]}}{d^{\text{IntPart}[p]}\,(d+e\,x+f\,x^2)^{\text{FracPart}[p]}}$

Rule 1.2.1.5.1.2: If $c\,d-a\,f==0\,\wedge\,b\,d-a\,e==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,q\notin\mathbb{Z}\,\wedge\,\frac{c}{f}\ngtr 0$, then

$$\int (a+b\,x+c\,x^2)^p\,(d+e\,x+f\,x^2)^q\,dx\,\to\,\frac{a^{\text{IntPart}[p]}\,(a+b\,x+c\,x^2)^{\text{FracPart}[p]}}{d^{\text{IntPart}[p]}\,(d+e\,x+f\,x^2)^{\text{FracPart}[p]}}\int (d+e\,x+f\,x^2)^{p+q}\,dx$$

Program code:

```
Int[(a_+b_.*x_+c_.*x_^2)^p_*(d_+e_.*x_+f_.*x_^2)^q_.,x_Symbol] :=
  a^IntPart[p]*(a+b*x+c*x^2)^FracPart[p]/(d^IntPart[p]*(d+e*x+f*x^2)^FracPart[p])*Int[(d+e*x+f*x^2)^(p+q),x] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && EqQ[c*d-a*f,0] && EqQ[b*d-a*e,0] && Not[IntegerQ[p]] && Not[IntegerQ[q]] && Not[GtQ[c/f,0]]
```

2: $\int (a+b\,x+c\,x^2)^p\,(d+e\,x+f\,x^2)^q\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x\frac{(a+b\,x+c\,x^2)^p}{(b+2\,c\,x)^{2\,p}}==0$

Basis: If $b^2-4\,a\,c==0$, then $\frac{(a+b\,x+c\,x^2)^p}{(b+2\,c\,x)^{2\,p}}==\frac{(a+b\,x+c\,x^2)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,(b+2\,c\,x)^{2\,\text{FracPart}[p]}}$

Rule 1.2.1.5.2: If $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$, then

$$\int \left(a+b\,x+c\,x^2\right)^p \left(d+e\,x+f\,x^2\right)^q\,dx \longrightarrow \frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,\left(b+2\,c\,x\right)^{2\,\text{FracPart}[p]}} \int \left(b+2\,c\,x\right)^{2\,p} \left(d+e\,x+f\,x^2\right)^q\,dx$$

Program code:

```
Int[(a_+b_.*x_+c_.*x_^2)^p_*(d_+e_.*x_+f_.*x_^2)^q_.,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x)^(2*FracPart[p]))*Int[(b+2*c*x)^(2*p)*(d+e*x+f*x^2)^q,x] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[(a_+b_.*x_+c_.*x_^2)^p_*(d_+f_.*x_^2)^q_.,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x)^(2*FracPart[p]))*Int[(b+2*c*x)^(2*p)*(d+f*x^2)^q,x] /;
FreeQ[{a,b,c,d,f,p,q},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

x. $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land c e-b f == 0$

1. $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land c e-b f == 0 \land \left(p \in \mathbb{Z} \lor -\frac{c}{b^2-4 a c}>0\right)$

1: $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land c e-b f == 0 \land \left(p \in \mathbb{Z} \lor -\frac{c}{b^2-4 a c}>0\right) \land \left(q \in \mathbb{Z} \lor -\frac{f}{e^2-4 d f}>0\right)$

Derivation: Algebraic simplification and integration by substitution

Basis: If $p \in \mathbb{Z} \lor -\frac{c}{b^2-4 a c}>0$, then $(a+b x+c x^2)^p == \frac{1}{2^{2 p} \left(-\frac{c}{b^2-4 a c}\right)^p} \left(1-\frac{(b+2 c x)^2}{b^2-4 a c}\right)^p$

Basis: If $c e-b f == 0 \land \left(q \in \mathbb{Z} \lor -\frac{f}{e^2-4 d f}>0\right)$, then $(d+e x+f x^2)^q == \frac{1}{2^{2 q} \left(-\frac{f}{e^2-4 d f}\right)^q} \left(1+\frac{e (b+2 c x)^2}{b (4 c d-b e)}\right)^q$

Rule 1.2.1.5.x.1.1: If
$b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land c e-b f == 0 \land \left(p \in \mathbb{Z} \lor -\frac{c}{b^2-4 a c}>0\right) \land \left(q \in \mathbb{Z} \lor -\frac{f}{e^2-4 d f}>0\right)$, then

$$\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx \to \frac{1}{2^{2 p+2 q} \left(-\frac{c}{b^2-4 a c}\right)^p \left(-\frac{f}{e^2-4 d f}\right)^q} \int \left(1-\frac{(b+2 c x)^2}{b^2-4 a c}\right)^p \left(1+\frac{e (b+2 c x)^2}{b (4 c d-b e)}\right)^q dx$$

$$\to \frac{1}{2^{2 p+2 q+1} c \left(-\frac{c}{b^2-4 a c}\right)^p \left(-\frac{f}{e^2-4 d f}\right)^q} \text{Subst}\left[\int \left(1-\frac{x^2}{b^2-4 a c}\right)^p \left(1+\frac{e x^2}{b (4 c d-b e)}\right)^q dx,\ x,\ b+2 c x\right]$$

Program code:

```
(* Int[(a_+b_.*x_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  1/(2^(2*p+2*q+1)*c*(-c/(b^2-4*a*c))^p*(-f/(e^2-4*d*f))^q)*
    Subst[Int[(1-x^2/(b^2-4*a*c))^p*(1+e*x^2/(b*(4*c*d-b*e)))^q,x],x,b+2*c*x] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && EqQ[c*e-b*f,0] &&
  (IntegerQ[p] || GtQ[-c/(b^2-4*a*c),0]) && (IntegerQ[q] || GtQ[-f/(e^2-4*d*f),0]) *)
```

2: $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $b^2-4 a c \neq 0 \wedge e^2-4 d f \neq 0 \wedge c e-b f == 0 \wedge \left(p \in \mathbb{Z} \vee -\frac{c}{b^2-4 a c}>0\right) \wedge \neg \left(q \in \mathbb{Z} \vee -\frac{f}{e^2-4 d f}>0\right)$

Derivation: Algebraic simplification, piecewise constant extraction, and integration by substitution

Basis: If $p \in \mathbb{Z} \vee -\frac{c}{b^2-4 a c}>0$, then $(a+b x+c x^2)^p == \frac{1}{2^{2 p} \left(-\frac{c}{b^2-4 a c}\right)^p} \left(1-\frac{(b+2 c x)^2}{b^2-4 a c}\right)^p$

Basis: $\partial_x \frac{F[x]^p}{(c\,F[x])^p} == 0$

Basis: If $c e-b f == 0$, then $-\frac{f (d+e x+f x^2)}{e^2-4 d f} == \frac{1}{2^2} \left(1+\frac{e (b+2 c x)^2}{b (4 c d-b e)}\right)$

Rule 1.2.1.5.x.1.2: If
$b^2-4 a c \neq 0 \wedge e^2-4 d f \neq 0 \wedge c e-b f == 0 \wedge \left(p \in \mathbb{Z} \vee -\frac{c}{b^2-4 a c}>0\right) \wedge \neg \left(q \in \mathbb{Z} \vee -\frac{f}{e^2-4 d f}>0\right)$, then

$$\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx \to \frac{(d+e x+f x^2)^q}{2^{2 p+2 q} \left(-\frac{c}{b^2-4 a c}\right)^p \left(-\frac{f (d+e x+f x^2)}{e^2-4 d f}\right)^q} \int \left(1-\frac{(b+2 c x)^2}{b^2-4 a c}\right)^p \left(1+\frac{e (b+2 c x)^2}{b (4 c d-b e)}\right)^q dx$$

$$\to \frac{(d+e x+f x^2)^q}{2^{2 p+2 q+1} c \left(-\frac{c}{b^2-4 a c}\right)^p \left(-\frac{f (d+e x+f x^2)}{e^2-4 d f}\right)^q} \text{Subst}\left[\int \left(1-\frac{x^2}{b^2-4 a c}\right)^p \left(1+\frac{e x^2}{b (4 c d-b e)}\right)^q dx,\ x,\ b+2 c x\right]$$

Program code:

```
(* Int[(a_+b_.*x_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  (d+e*x+f*x^2)^q/(2^(2*p+2*q+1)*c*(-c/(b^2-4*a*c))^p*(-f*(d+e*x+f*x^2)/(e^2-4*d*f))^q)*
    Subst[Int[(1-x^2/(b^2-4*a*c))^p*(1+e*x^2/(b*(4*c*d-b*e)))^q,x],x,b+2*c*x] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && EqQ[c*e-b*f,0] &&
  (IntegerQ[p] || GtQ[-c/(b^2-4*a*c),0]) && Not[IntegerQ[q] || GtQ[-f/(e^2-4*d*f),0]] *)
```

2: $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land c e-b f == 0 \land \neg \left(p \in \mathbb{Z} \lor -\frac{c}{b^2-4 a c}>0\right) \land \neg \left(q \in \mathbb{Z} \lor -\frac{f}{e^2-4 d f}>0\right)$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{F[x]^p}{(c F[x])^p} == 0$

Basis: $-\frac{c (a+b x+c x^2)}{b^2-4 a c} == \frac{1}{2^2} \left(1-\frac{(b+2 c x)^2}{b^2-4 a c}\right)$

Basis: If $c e-b f == 0$, then $-\frac{f (d+e x+f x^2)}{e^2-4 d f} == \frac{1}{2^2} \left(1+\frac{e (b+2 c x)^2}{b (4 c d-b e)}\right)$

Rule 1.2.1.5.x.2: If $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land c e-b f == 0$, then

$$\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx \to \frac{(a+b x+c x^2)^p (d+e x+f x^2)^q}{2^{2 p+2 q} \left(-\frac{c (a+b x+c x^2)}{b^2-4 a c}\right)^p \left(-\frac{f (d+e x+f x^2)}{e^2-4 d f}\right)^q} \int \left(1-\frac{(b+2 c x)^2}{b^2-4 a c}\right)^p \left(1+\frac{e (b+2 c x)^2}{b (4 c d-b e)}\right)^q dx$$

$$\to \frac{(a+b x+c x^2)^p (d+e x+f x^2)^q}{2^{2 p+2 q+1} c \left(-\frac{c (a+b x+c x^2)}{b^2-4 a c}\right)^p \left(-\frac{f (d+e x+f x^2)}{e^2-4 d f}\right)^q} \text{Subst}\left[\int \left(1-\frac{x^2}{b^2-4 a c}\right)^p \left(1+\frac{e x^2}{b (4 c d-b e)}\right)^q dx, x, b+2 c x\right]$$

Program code:

```
(* Int[(a_+b_.*x_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  (a+b*x+c*x^2)^p*(d+e*x+f*x^2)^q/(2^(2*p+2*q+1)*c*(-c*(a+b*x+c*x^2)/(b^2-4*a*c))^p*(-f*(d+e*x+f*x^2)/(e^2-4*d*f))^q)*
    Subst[Int[(1-x^2/(b^2-4*a*c))^p*(1+e*x^2/(b*(4*c*d-b*e)))^q,x],x,b+2*c*x] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && EqQ[c*e-b*f,0] *)
```

4. $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land p<-1$

1: $\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land p<-1 \land q>0$

Derivation: Nondegenerate biquadratic recurrence 1 with $A \to 1,\ B \to 0,\ C \to 0$

Rule 1.2.1.5.4.1: If $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land p<-1 \land q>0$, then

$$\int (a+b x+c x^2)^p (d+e x+f x^2)^q \, dx \longrightarrow \frac{(b+2 c x) (a+b x+c x^2)^{p+1} (d+e x+f x^2)^q}{(b^2-4 a c) (p+1)} -$$

$$\frac{1}{(b^2-4 a c) (p+1)} \int (a+b x+c x^2)^{p+1} (d+e x+f x^2)^{q-1} \left(2 c d (2 p+3)+b e q+(2 b f q+2 c e (2 p+q+3)) x+2 c f (2 p+2 q+3) x^2\right) dx$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^(p+1)*(d+e*x+f*x^2)^q/((b^2-4*a*c)*(p+1)) -
  (1/((b^2-4*a*c)*(p+1)))*
    Int[(a+b*x+c*x^2)^(p+1)*(d+e*x+f*x^2)^(q-1)*
      Simp[2*c*d*(2*p+3)+b*e*q+(2*b*f*q+2*c*e*(2*p+q+3))*x+2*c*f*(2*p+2*q+3)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && LtQ[p,-1] && GtQ[q,0] && Not[IGtQ[q,0]]
```

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_*(d_.+f_.*x_^2)^q_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^(p+1)*(d+f*x^2)^q/((b^2-4*a*c)*(p+1)) -
  (1/((b^2-4*a*c)*(p+1)))*
    Int[(a+b*x+c*x^2)^(p+1)*(d+f*x^2)^(q-1)*
      Simp[2*c*d*(2*p+3)+(2*b*f*q)*x+2*c*f*(2*p+2*q+3)*x^2,x],x] /;
FreeQ[{a,b,c,d,f},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[q,0] && Not[IGtQ[q,0]]
```

```
Int[(a_.+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  (2*c*x)*(a+c*x^2)^(p+1)*(d+e*x+f*x^2)^q/((-4*a*c)*(p+1)) -
  (1/((-4*a*c)*(p+1)))*
    Int[(a+c*x^2)^(p+1)*(d+e*x+f*x^2)^(q-1)*
      Simp[2*c*d*(2*p+3)+(2*c*e*(2*p+q+3))*x+2*c*f*(2*p+2*q+3)*x^2,x],x] /;
FreeQ[{a,c,d,e,f},x] && NeQ[e^2-4*d*f] && LtQ[p,-1] && GtQ[q,0] && Not[IGtQ[q,0]]
```

2: $\int (a+bx+cx^2)^p (d+ex+fx^2)^q\,dx$ when $b^2-4ac \neq 0 \land e^2-4df \neq 0 \land p < -1 \land q \not> 0 \land (cd-af)^2-(bd-ae)(ce-bf) \neq 0$

Derivation: Nondegenerate biquadratic recurrence 3 with $A \to 1,\ B \to 0,\ C \to 0$

Rule 1.2.1.5.4.2: If
$b^2-4ac \neq 0 \land e^2-4df \neq 0 \land p < -1 \land q \not> 0 \land (cd-af)^2-(bd-ae)(ce-bf) \neq 0$, then

$$\begin{aligned}
&\int (a+bx+cx^2)^p (d+ex+fx^2)^q\,dx \longrightarrow \\
&\Big(\big(\big(2ac^2e-b^2ce+b^3f+bc(cd-3af)+c\,(2c^2d+b^2f-c(be+2af))\,x\big)(a+bx+cx^2)^{p+1}(d+ex+fx^2)^{q+1}\big)\Big/ \\
&\big((b^2-4ac)\,((cd-af)^2-(bd-ae)(ce-bf))\,(p+1)\big)\Big) - \\
&\frac{1}{(b^2-4ac)\,((cd-af)^2-(bd-ae)(ce-bf))\,(p+1)}\int (a+bx+cx^2)^{p+1}(d+ex+fx^2)^q\,\cdot \\
&\quad\big(2c\,((cd-af)^2-(bd-ae)(ce-bf))\,(p+1)- \\
&\quad(2c^2d+b^2f-c(be+2af))\,(af(p+1)-cd(p+2))- \\
&\quad e\,(b^2ce-2ac^2e-b^3f-bc(cd-3af))\,(p+q+2)+ \\
&\quad(2f\,(2ac^2e-b^2ce+b^3f+bc(cd-3af))\,(p+q+2)-(2c^2d+b^2f-c(be+2af))\,(bf(p+1)-ce(2p+q+4)))\,x+ \\
&\quad cf\,(2c^2d+b^2f-c(be+2af))\,(2p+2q+5)\,x^2\big)\,dx
\end{aligned}$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  (2*a*c^2*e-b^2*c*e+b^3*f+b*c*(c*d-3*a*f)+c*(2*c^2*d+b^2*f-c*(b*e+2*a*f))*x)*(a+b*x+c*x^2)^(p+1)*(d+e*x+f*x^2)^(q+1)/
    ((b^2-4*a*c)*((c*d-a*f)^2-(b*d-a*e)*(c*e-b*f))*(p+1)) -
  (1/((b^2-4*a*c)*((c*d-a*f)^2-(b*d-a*e)*(c*e-b*f))*(p+1)))*
    Int[(a+b*x+c*x^2)^(p+1)*(d+e*x+f*x^2)^q*
      Simp[2*c*((c*d-a*f)^2-(b*d-a*e)*(c*e-b*f))*(p+1)-
        (2*c^2*d+b^2*f-c*(b*e+2*a*f))*(a*f*(p+1)-c*d*(p+2))-
        e*(b^2*c*e-2*a*c^2*e-b^3*f-b*c*(c*d-3*a*f))*(p+q+2)+
       (2*f*(2*a*c^2*e-b^2*c*e+b^3*f+b*c*(c*d-3*a*f))*(p+q+2)-(2*c^2*d+b^2*f-c*(b*e+2*a*f))*(b*f*(p+1)-c*e*(2*p+q+4)))*x+
       c*f*(2*c^2*d+b^2*f-c*(b*e+2*a*f))*(2*p+2*q+5)*x^2,x],x]/;
FreeQ[{a,b,c,d,e,f,q},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && LtQ[p,-1] &&
  NeQ[(c*d-a*f)^2-(b*d-a*e)*(c*e-b*f),0] && Not[Not[IntegerQ[p]] && ILtQ[q,-1]] && Not[IGtQ[q,0]]
```

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_*(d_.+f_.*x_^2)^q_,x_Symbol] :=
  (b^3*f+b*c*(c*d-3*a*f)+c*(2*c^2*d+b^2*f-c*(2*a*f))*x)*(a+b*x+c*x^2)^(p+1)*(d+f*x^2)^(q+1)/
    ((b^2-4*a*c)*(b^2*d*f+(c*d-a*f)^2)*(p+1)) -
  (1/((b^2-4*a*c)*(b^2*d*f+(c*d-a*f)^2)*(p+1)))*
    Int[(a+b*x+c*x^2)^(p+1)*(d+f*x^2)^q*
      Simp[2*c*(b^2*d*f+(c*d-a*f)^2)*(p+1)-
        (2*c^2*d+b^2*f-c*(2*a*f))*(a*f*(p+1)-c*d*(p+2))+
       (2*f*(b^3*f+b*c*(c*d-3*a*f))*(p+q+2)-(2*c^2*d+b^2*f-c*(2*a*f))*(b*f*(p+1)))*x+
       c*f*(2*c^2*d+b^2*f-c*(2*a*f))*(2*p+2*q+5)*x^2,x],x]/;
FreeQ[{a,b,c,d,f,q},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && NeQ[b^2*d*f+(c*d-a*f)^2,0] &&
  Not[Not[IntegerQ[p]] && ILtQ[q,-1]] && Not[IGtQ[q,0]]
```

```
Int[(a_.+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  (2*a*c^2*e+c*(2*c^2*d-c*(2*a*f))*x)*(a+c*x^2)^(p+1)*(d+e*x+f*x^2)^(q+1)/
    ((-4*a*c)*(a*c*e^2+(c*d-a*f)^2)*(p+1)) -
  (1/((-4*a*c)*(a*c*e^2+(c*d-a*f)^2)*(p+1)))*
    Int[(a+c*x^2)^(p+1)*(d+e*x+f*x^2)^q*
      Simp[2*c*((c*d-a*f)^2-(-a*e)*(c*e))*(p+1)-(2*c^2*d-c*(2*a*f))*(a*f*(p+1)-c*d*(p+2))-e*(-2*a*c^2*e)*(p+q+2)+
       (2*f*(2*a*c^2*e)*(p+q+2)-(2*c^2*d-c*(2*a*f))*(-c*e*(2*p+q+4)))*x+
       c*f*(2*c^2*d-c*(2*a*f))*(2*p+2*q+5)*x^2,x],x]/;
FreeQ[{a,c,d,e,f,q},x] && NeQ[e^2-4*d*f,0] && LtQ[p,-1] && NeQ[a*c*e^2+(c*d-a*f)^2,0] &&
  Not[Not[IntegerQ[p]] && ILtQ[q,-1]] && Not[IGtQ[q,0]]
```

5: $\int \left(a+b x+c x^2\right)^p \left(d+e x+f x^2\right)^q \, dx$ when $b^2-4 a c\neq 0 \land e^2-4 d f\neq 0 \land p>1 \land p+q\neq 0 \land 2 p+2 q+1\neq 0$

Derivation: Nondegenerate biquadratic recurrence 2 with $A\to a,\ B\to b,\ C\to c,\ p\to p-1$

Rule 1.2.1.5.5: If $b^2-4 a c\neq 0 \land e^2-4 d f\neq 0 \land p>1 \land p+q\neq 0 \land 2 p+2 q+1\neq 0$, then

$$\int \left(a+b x+c x^2\right)^p \left(d+e x+f x^2\right)^q \, dx \longrightarrow$$

$$\left(\left(\left(b f (3 p+2 q)-c e (2 p+q)+2 c f (p+q) x\right) \left(a+b x+c x^2\right)^{p-1} \left(d+e x+f x^2\right)^{q+1}\right)\Big/\left(2 f^2 (p+q) (2 p+2 q+1)\right)\right)-$$

$$\frac{1}{2 f^2 (p+q) (2 p+2 q+1)} \int \left(a+b x+c x^2\right)^{p-2} \left(d+e x+f x^2\right)^q \cdot$$

$$\Big(\left(b d-a e\right) \left(c e-b f\right) (1-p) (2 p+q)-(p+q) \left(b^2 d f (1-p)-a \left(f \left(b e-2 a f\right) (2 p+2 q+1)+c \left(2 d f-e^2 (2 p+q)\right)\right)\right)+$$

$$\left(2\left(c d-a f\right)\left(c e-b f\right)(1-p)(2 p+q)-(p+q)\left(\left(b^2-4 a c\right) e f (1-p)+b\left(c\left(e^2-4 d f\right)(2 p+q)+f\left(2 c d-b e+2 a f\right)(2 p+2 q+1)\right)\right)\right) x+\left(\left(c e-b f\right)^2(1-p) p+c (p+q)\left(f\left(b e-2 a f\right)(4 p+2 q-1)-c\left(2 d f (1-2 p)+e^2 (3 p+q-1)\right)\right)\right) x^2\Big) dx$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  (b*f*(3*p+2*q)-c*e*(2*p+q)+2*c*f*(p+q)*x)*(a+b*x+c*x^2)^(p-1)*(d+e*x+f*x^2)^(q+1)/(2*f^2*(p+q)*(2*p+2*q+1)) -
  1/(2*f^2*(p+q)*(2*p+2*q+1))*
    Int[(a+b*x+c*x^2)^(p-2)*(d+e*x+f*x^2)^q*
      Simp[(b*d-a*e)*(c*e-b*f)*(1-p)*(2*p+q)-
        (p+q)*(b^2*d*f*(1-p)-a*(f*(b*e-2*a*f)*(2*p+2*q+1)+c*(2*d*f-e^2*(2*p+q))))+
        (2*(c*d-a*f)*(c*e-b*f)*(1-p)*(2*p+q)-
          (p+q)*((b^2-4*a*c)*e*f*(1-p)+b*(c*(e^2-4*d*f)*(2*p+q)+f*(2*c*d-b*e+2*a*f)*(2*p+2*q+1))))*x+
        ((c*e-b*f)^2*(1-p)*p+c*(p+q)*(f*(b*e-2*a*f)*(4*p+2*q-1)-c*(2*d*f*(1-2*p)+e^2*(3*p+q-1))))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,q},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && GtQ[p,1] &&
  NeQ[p+q,0] && NeQ[2*p+2*q+1,0] && Not[IGtQ[p,0]] && Not[IGtQ[q,0]]
```

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_*(d_.+f_.*x_^2)^q_,x_Symbol] :=
  (b*(3*p+2*q)+2*c*(p+q)*x)*(a+b*x+c*x^2)^(p-1)*(d+f*x^2)^(q+1)/(2*f*(p+q)*(2*p+2*q+1)) -
  1/(2*f*(p+q)*(2*p+2*q+1))*
    Int[(a+b*x+c*x^2)^(p-2)*(d+f*x^2)^q*
      Simp[b^2*d*(p-1)*(2*p+q)-(p+q)*(b^2*d*(1-p)-2*a*(c*d-a*f*(2*p+2*q+1)))-
        (2*b*(c*d-a*f)*(1-p)*(2*p+q)-2*(p+q)*b*(2*c*d*(2*p+q)-(c*d+a*f)*(2*p+2*q+1)))*x+
        (b^2*f*p*(1-p)+2*c*(p+q)*(c*d*(2*p-1)-a*f*(4*p+2*q-1)))*x^2,x],x] /;
FreeQ[{a,b,c,d,f,q},x] && NeQ[b^2-4*a*c,0] && GtQ[p,1] && NeQ[p+q,0] && NeQ[2*p+2*q+1,0] && Not[IGtQ[p,0]] && Not[IGtQ[q,0]]
```

```
Int[(a_.+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  -c*(e*(2*p+q)-2*f*(p+q)*x)*(a+c*x^2)^(p-1)*(d+e*x+f*x^2)^(q+1)/(2*f^2*(p+q)*(2*p+2*q+1)) -
  1/(2*f^2*(p+q)*(2*p+2*q+1))*
    Int[(a+c*x^2)^(p-2)*(d+e*x+f*x^2)^q*
      Simp[-a*c*e^2*(1-p)*(2*p+q)+a*(p+q)*(-2*a*f^2*(2*p+2*q+1)+c*(2*d*f-e^2*(2*p+q)))+
        (2*(c*d-a*f)*(c*e)*(1-p)*(2*p+q)+4*a*c*e*f*(1-p)*(p+q))*x+
        (p*c^2*e^2*(1-p)-c*(p+q)*(2*a*f^2*(4*p+2*q-1)+c*(2*d*f*(1-2*p)+e^2*(3*p+q-1))))*x^2,x],x] /;
FreeQ[{a,c,d,e,f,q},x] && NeQ[e^2-4*d*f,0] && GtQ[p,1] && NeQ[p+q,0] && NeQ[2*p+2*q+1,0] && Not[IGtQ[p,0]] && Not[IGtQ[q,0]]
```

6: $\int \frac{1}{\left(a+b x+c x^2\right)\left(d+e x+f x^2\right)}\,dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0 \land c^2 d^2-b c d e+a c e^2+b^2 d f-2 a c d f-a b e f+a^2 f^2 \neq 0$

Derivation: Algebraic expansion

Basis: Let $q = c^2 d^2-b c d e+a c e^2+b^2 d f-2 a c d f-a b e f+a^2 f^2$, then

$$\frac{1}{\left(a+b x+c x^2\right)\left(d+e x+f x^2\right)} == \frac{c^2 d-b c e+b^2 f-a c f-\left(c^2 e-b c f\right) x}{q\left(a+b x+c x^2\right)}+\frac{c e^2-c d f-b e f+a f^2+\left(c e f-b f^2\right) x}{q\left(d+e x+f x^2\right)}$$

Rule 1.2.1.5.6: If $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0$, let
$q = c^2 d^2-b c d e+a c e^2+b^2 d f-2 a c d f-a b e f+a^2 f^2$, if $q \neq 0$, then

$$\int \frac{1}{\left(a+b x+c x^2\right)\left(d+e x+f x^2\right)}\,dx \to \frac{1}{q}\int \frac{c^2 d-b c e+b^2 f-a c f-\left(c^2 e-b c f\right) x}{a+b x+c x^2}\,dx+\frac{1}{q}\int \frac{c e^2-c d f-b e f+a f^2+\left(c e f-b f^2\right) x}{d+e x+f x^2}\,dx$$

Program code:

```
Int[1/((a_+b_.*x_+c_.*x_^2)*(d_+e_.*x_+f_.*x_^2)),x_Symbol] :=
  With[{q=c^2*d^2-b*c*d*e+a*c*e^2+b^2*d*f-2*a*c*d*f-a*b*e*f+a^2*f^2},
  1/q*Int[(c^2*d-b*c*e+b^2*f-a*c*f-(c^2*e-b*c*f)*x)/(a+b*x+c*x^2),x] +
  1/q*Int[(c*e^2-c*d*f-b*e*f+a*f^2+(c*e*f-b*f^2)*x)/(d+e*x+f*x^2),x] /;
 NeQ[q,0]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0]
```

```
Int[1/((a_+b_.*x_+c_.*x_^2)*(d_+f_.*x_^2)),x_Symbol] :=
  With[{q=c^2*d^2+b^2*d*f-2*a*c*d*f+a^2*f^2},
  1/q*Int[(c^2*d+b^2*f-a*c*f+b*c*f*x)/(a+b*x+c*x^2),x] -
  1/q*Int[(c*d*f-a*f^2+b*f^2*x)/(d+f*x^2),x] /;
 NeQ[q,0]] /;
FreeQ[{a,b,c,d,f},x] && NeQ[b^2-4*a*c,0]
```

7. $\int \frac{1}{(a+b x+c x^2)\sqrt{d+e x+f x^2}}\,dx$ when $b^2-4ac\neq 0 \land e^2-4df\neq 0$

1: $\int \frac{1}{(a+b x+c x^2)\sqrt{d+e x+f x^2}}\,dx$ when $b^2-4ac\neq 0 \land e^2-4df\neq 0 \land c e-b f == 0$

Reference: G&R 2.252.3b

Derivation: Integration by substitution

Basis: If $c e-b f == 0$, then

$$\frac{1}{(a+b x+c x^2)\sqrt{d+e x+f x^2}} == -2 e \,\text{Subst}\left[\frac{1}{e (b e-4 a f)-(b d-a e) x^2}, x, \frac{e+2 f x}{\sqrt{d+e x+f x^2}}\right] \partial_x \frac{e+2 f x}{\sqrt{d+e x+f x^2}}$$

Rule 1.2.1.5.7.1: If $b^2-4ac\neq 0 \land e^2-4df\neq 0 \land c e-b f == 0$, then

$$\int \frac{1}{(a+b x+c x^2)\sqrt{d+e x+f x^2}}\,dx \rightarrow -2 e \,\text{Subst}\left[\int \frac{1}{e (b e-4 a f)-(b d-a e) x^2}\,dx, x, \frac{e+2 f x}{\sqrt{d+e x+f x^2}}\right]$$

Program code:

```
Int[1/((a_+b_.*x_+c_.*x_^2)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  -2*e*Subst[Int[1/(e*(b*e-4*a*f)-(b*d-a*e)*x^2),x],x,(e+2*f*x)/Sqrt[d+e*x+f*x^2]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && EqQ[c*e-b*f,0]
```

2. $\int \frac{1}{(a+b x+c x^2)\sqrt{d+e x+f x^2}}\,dx$ when $b^2-4 a c\neq 0 \land e^2-4 d f\neq 0 \land c e-b f\neq 0$

x: $\int \frac{1}{(a+b x+c x^2)\sqrt{d+e x+f x^2}}\,dx$ when $b^2-4 a c\neq 0 \land e^2-4 d f\neq 0 \land c e-b f\neq 0 \land b^2-4 a c<0$

Reference: G&R 2.252.3a

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{1}{\sqrt{d+e x+f x^2}}\,(c d-a f+c f k+(c e-b f)\,x)\sqrt{(d+e x+f x^2)\left(\frac{c f k}{c d-a f+c f k+(c e-b f)\,x}\right)^2}==0$

Basis: Let $k\to\sqrt{\left(\frac{a}{c}-\frac{d}{f}\right)^2+\left(\frac{b}{c}-\frac{e}{f}\right)\left(\frac{b d}{c f}-\frac{a e}{c f}\right)}$, then

$$1\Big/\left((a+b x+c x^2)\,(c d-a f+c f k+(c e-b f)\,x)\sqrt{(d+e x+f x^2)\left(\frac{c f k}{c d-a f+c f k+(c e-b f)\,x}\right)^2}\right)==$$

$$-\frac{2}{c}\,\text{Subst}\Bigg[(1-x)\Big/\left(\left(b d-a e-b f k-\frac{(c d-a f-c f k)^2}{c e-b f}+\left(b d-a e+b f k-\frac{(a f-c d-c f k)^2}{c e-b f}\right)x^2\right)\right.$$
$$\left.\sqrt{\left(-f\left(\frac{(b d-a e-c e k)}{c e-b f}-\frac{(c d-a f-c f k)^2}{(c e-b f)^2}\right)-f\left(\frac{b d-a e+c e k}{c e-b f}-\frac{(a f-c d-c f k)^2}{(c e-b f)^2}\right)x^2\right)}\right),$$
$$x,\ \frac{c d-a f-c f k+(c e-b f)\,x}{c d-a f+c f k+(c e-b f)\,x}\Bigg]\ \partial_x \frac{c d-a f-c f k+(c e-b f)\,x}{c d-a f+c f k+(c e-b f)\,x}$$

Rule 1.2.1.5.7.2.x: If $b^2-4 a c\neq 0 \land e^2-4 d f\neq 0 \land c e-b f\neq 0 \land b^2-4 a c<0$, then

$$\int \frac{1}{(a+b x+c x^2)\sqrt{d+e x+f x^2}}\,dx \longrightarrow$$

$$\frac{1}{\sqrt{d+e\,x+f\,x^2}}\left(c\,d-a\,f+c\,f\,k+\left(c\,e-b\,f\right)\,x\right)\sqrt{\left(d+e\,x+f\,x^2\right)\left(\frac{c\,f\,k}{c\,d-a\,f+c\,f\,k+\left(c\,e-b\,f\right)\,x}\right)^2}$$

$$\int\left(1\Big/\left(\left(a+b\,x+c\,x^2\right)\left(c\,d-a\,f+c\,f\,k+\left(c\,e-b\,f\right)\,x\right)\sqrt{\left(d+e\,x+f\,x^2\right)\left(\frac{c\,f\,k}{c\,d-a\,f+c\,f\,k+\left(c\,e-b\,f\right)\,x}\right)^2}\right)\right)dx\,\rightarrow$$

$$-\left(\left(2\left(c\,d-a\,f+c\,f\,k+\left(c\,e-b\,f\right)\,x\right)\sqrt{\left(d+e\,x+f\,x^2\right)\left(\frac{c\,f\,k}{c\,d-a\,f+c\,f\,k+\left(c\,e-b\,f\right)\,x}\right)^2}\right)\Big/\left(c\,\sqrt{d+e\,x+f\,x^2}\right)\right)\cdot$$

$$\text{Subst}\Big[\int\left(1-x\right)\Big/\left(\left(b\,d-a\,e-b\,f\,k-\frac{\left(c\,d-a\,f-c\,f\,k\right)^2}{c\,e-b\,f}+\left(b\,d-a\,e+b\,f\,k-\frac{\left(a\,f-c\,d-c\,f\,k\right)^2}{c\,e-b\,f}\right)x^2\right)\right.$$
$$\left.\sqrt{\left(-f\left(\frac{\left(b\,d-a\,e-c\,e\,k\right)}{c\,e-b\,f}-\frac{\left(c\,d-a\,f-c\,f\,k\right)^2}{\left(c\,e-b\,f\right)^2}\right)-f\left(\frac{b\,d-a\,e+c\,e\,k}{c\,e-b\,f}-\frac{\left(a\,f-c\,d-c\,f\,k\right)^2}{\left(c\,e-b\,f\right)^2}\right)x^2\right)}\right)dx,\,x,\,\frac{c\,d-a\,f-c\,f\,k+\left(c\,e-b\,f\right)\,x}{c\,d-a\,f+c\,f\,k+\left(c\,e-b\,f\right)\,x}\Big]$$

Program code:

```
(* Int[1/((a_+b_.*x_+c_.*x_^2)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{k=Rt[(a/c-d/f)^2+(b/c-e/f)*(b*d/(c*f)-a*e/(c*f)),2]},
  -2*(c*d-a*f+c*f*k+(c*e-b*f)*x)*Sqrt[(d+e*x+f*x^2)*((c*f*k)/(c*d-a*f+c*f*k+(c*e-b*f)*x))^2]/(c*Sqrt[d+e*x+f*x^2])*
    Subst[Int[(1-x)/(
      (b*d-a*e-b*f*k-(c*d-a*f-c*f*k)^2/(c*e-b*f)+(b*d-a*e+b*f*k-(a*f-c*d-c*f*k)^2/(c*e-b*f))*x^2)*
      Sqrt[-f*((b*d-a*e-c*e*k)/(c*e-b*f)-(c*d-a*f-c*f*k)^2/(c*e-b*f)^2)-f*((b*d-a*e+c*e*k)/(c*e-b*f)-(a*f-c*d-c*f*k)^2/(c*e-b*f)^2)*
        (c*d-a*f-c*f*k+(c*e-b*f)*x)/(c*d-a*f+c*f*k+(c*e-b*f)*x)]] /;
FreeQ[{a,b,c,d,e,f},x] && RationalQ[a,b,c,d,e,f] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && NeQ[c*e-b*f,0] && LtQ[b^2-4*a*c,0] *)
```

```
(* Int[1/((a_+c_.*x_^2)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{k=Rt[(a/c-d/f)^2+a*e^2/(c*f^2),2]},
  -2*(c*d-a*f+c*f*k+c*e*x)*Sqrt[(d+e*x+f*x^2)*((c*f*k)/(c*d-a*f+c*f*k+c*e*x))^2]/(c*Sqrt[d+e*x+f*x^2])*
    Subst[Int[(1-x)/(
      (-a*e-(c*d-a*f-c*f*k)^2/(c*e)+(-a*e-(a*f-c*d-c*f*k)^2/(c*e))*x^2)*
      Sqrt[-f*((-a*e-c*e*k)/(c*e)-(c*d-a*f-c*f*k)^2/(c*e)^2)-f*((-a*e+c*e*k)/(c*e)-(a*f-c*d-c*f*k)^2/(c*e)^2)*x^2]),x],x,
        (c*d-a*f-c*f*k+(c*e)*x)/(c*d-a*f+c*f*k+(c*e)*x)]] /;
FreeQ[{a,c,d,e,f},x] && RationalQ[a,c,d,e,f] && NeQ[e^2-4*d*f,0] && LtQ[-a*c,0] *)
```

```
(* Int[1/((a_+b_.*x_+c_.*x_^2)*Sqrt[d_.+f_.*x_^2]),x_Symbol] :=
  With[{k=Rt[(a/c-d/f)^2+b^2*d/(c^2*f),2]},
  -2*(c*d-a*f+c*f*k-b*f*x)*Sqrt[(d+f*x^2)*((c*f*k)/(c*d-a*f+c*f*k-b*f*x))^2]/(c*Sqrt[d+f*x^2])*
    Subst[Int[(1-x)/(
      (b*d-b*f*k+(c*d-a*f-c*f*k)^2/(b*f)+(b*d+b*f*k+(a*f-c*d-c*f*k)^2/(b*f))*x^2)*
      Sqrt[-f*(-d/f-(c*d-a*f-c*f*k)^2/(b*f)^2)-f*(-d/f-(a*f-c*d-c*f*k)^2/(b*f)^2)*x^2]),x],x,
        (c*d-a*f-c*f*k+(-b*f)*x)/(c*d-a*f+c*f*k+(-b*f)*x)]] /;
FreeQ[{a,b,c,d,f},x] && RationalQ[a,b,c,d,f] && NeQ[b^2-4*a*c,0] && LtQ[b^2-4*a*c,0] *)
```

1: $\int \frac{1}{\left(a+b\,x+c\,x^2\right)\sqrt{d+e\,x+f\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,e^2-4\,d\,f\neq 0\,\land\,c\,e-b\,f\neq 0\,\land\,b^2-4\,a\,c>0$

Derivation: Algebraic expansion

Basis: Let $q=\sqrt{b^2-4\,a\,c}$, then $\frac{1}{a+b\,x+c\,x^2}==\frac{2\,c}{q}\,\frac{1}{(b-q+2\,c\,x)}-\frac{2\,c}{q}\,\frac{1}{(b+q+2\,c\,x)}$

Rule 1.2.1.5.7.2.1: If $b^2-4\,a\,c\neq 0\,\land\,e^2-4\,d\,f\neq 0\,\land\,c\,e-b\,f\neq 0\,\land\,b^2-4\,a\,c>0$, let $q=\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{1}{\left(a+b\,x+c\,x^2\right)\sqrt{d+e\,x+f\,x^2}}\,dx\,\rightarrow\,\frac{2\,c}{q}\int \frac{1}{(b-q+2\,c\,x)\sqrt{d+e\,x+f\,x^2}}\,dx-\frac{2\,c}{q}\int \frac{1}{(b+q+2\,c\,x)\sqrt{d+e\,x+f\,x^2}}\,dx$$

Program code:

```
Int[1/((a_+b_.*x_+c_.*x_^2)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/((b-q+2*c*x)*Sqrt[d+e*x+f*x^2]),x] -
  2*c/q*Int[1/((b+q+2*c*x)*Sqrt[d+e*x+f*x^2]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && NeQ[c*e-b*f,0] && PosQ[b^2-4*a*c]
```

```
Int[1/((a_+c_.*x_^2)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  1/2*Int[1/((a-Rt[-a*c,2]*x)*Sqrt[d+e*x+f*x^2]),x] +
  1/2*Int[1/((a+Rt[-a*c,2]*x)*Sqrt[d+e*x+f*x^2]),x] /;
FreeQ[{a,c,d,e,f},x] && NeQ[e^2-4*d*f,0] && PosQ[-a*c]
```

```
Int[1/((a_+b_.*x_+c_.*x_^2)*Sqrt[d_+f_.*x_^2]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/((b-q+2*c*x)*Sqrt[d+f*x^2]),x] -
  2*c/q*Int[1/((b+q+2*c*x)*Sqrt[d+f*x^2]),x]] /;
FreeQ[{a,b,c,d,f},x] && NeQ[b^2-4*a*c,0] && PosQ[b^2-4*a*c]
```

2: $\int \frac{1}{\left(a+b\,x+c\,x^2\right)\sqrt{d+e\,x+f\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,e^2-4\,d\,f\neq 0\,\wedge\,c\,e-b\,f\neq 0\,\wedge\,b^2-4\,a\,c\not>0$

Derivation: Algebraic expansion

Note: If $b^2-4\,a\,c==\frac{1}{(c\,e-b\,f)^2}\left(\left(b\,(c\,e-b\,f)-2\,c\,(c\,d-a\,f)\right)^2-4\,c^2\left((c\,d-a\,f)^2-(b\,d-a\,e)\,(c\,e-b\,f)\right)\right)<0$, then $(c\,d-a\,f)^2-(b\,d-a\,e)\,(c\,e-b\,f)>0$ (noted by Martin Welz on sci.math.symbolic on 24 May 2015).

Note: Resulting integrands are of the form $\frac{g+h\,x}{\left(a+b\,x+c\,x^2\right)\sqrt{d+e\,x+f\,x^2}}$ where $h^2\,(b\,d-a\,e)-2\,g\,h\,(c\,d-a\,f)+g^2\,(c\,e-b\,f)==0$ for which there is a rule.

Rule 1.2.1.5.7.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,e^2-4\,d\,f\neq 0\,\wedge\,c\,e-b\,f\neq 0\,\wedge\,b^2-4\,a\,c\not>0$, let $q\to\sqrt{(c\,d-a\,f)^2-(b\,d-a\,e)\,(c\,e-b\,f)}$, then

$$\int \frac{1}{\left(a+b\,x+c\,x^2\right)\sqrt{d+e\,x+f\,x^2}}\,dx \to \frac{1}{2\,q}\int \frac{c\,d-a\,f+q+(c\,e-b\,f)\,x}{\left(a+b\,x+c\,x^2\right)\sqrt{d+e\,x+f\,x^2}}\,dx-\frac{1}{2\,q}\int \frac{c\,d-a\,f-q+(c\,e-b\,f)\,x}{\left(a+b\,x+c\,x^2\right)\sqrt{d+e\,x+f\,x^2}}\,dx$$

Program code:

```
Int[1/((a_.+b_.*x_+c_.*x_^2)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{q=Rt[(c*d-a*f)^2-(b*d-a*e)*(c*e-b*f),2]},
  1/(2*q)*Int[(c*d-a*f+q+(c*e-b*f)*x)/((a+b*x+c*x^2)*Sqrt[d+e*x+f*x^2]),x] -
  1/(2*q)*Int[(c*d-a*f-q+(c*e-b*f)*x)/((a+b*x+c*x^2)*Sqrt[d+e*x+f*x^2]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0] && NeQ[c*e-b*f,0] && NegQ[b^2-4*a*c]
```

```
Int[1/((a_.+c_.*x_^2)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{q=Rt[(c*d-a*f)^2+a*c*e^2,2]},
  1/(2*q)*Int[(c*d-a*f+q+c*e*x)/((a+c*x^2)*Sqrt[d+e*x+f*x^2]),x] -
  1/(2*q)*Int[(c*d-a*f-q+c*e*x)/((a+c*x^2)*Sqrt[d+e*x+f*x^2]),x]] /;
FreeQ[{a,c,d,e,f},x] && NeQ[e^2-4*d*f,0] && NegQ[-a*c]
```

```
Int[1/((a_.+b_.*x_+c_.*x_^2)*Sqrt[d_.+f_.*x_^2]),x_Symbol] :=
  With[{q=Rt[(c*d-a*f)^2+b^2*d*f,2]},
  1/(2*q)*Int[(c*d-a*f+q+(-b*f)*x)/((a+b*x+c*x^2)*Sqrt[d+f*x^2]),x] -
  1/(2*q)*Int[(c*d-a*f-q+(-b*f)*x)/((a+b*x+c*x^2)*Sqrt[d+f*x^2]),x]] /;
FreeQ[{a,b,c,d,f},x] && NeQ[b^2-4*a*c,0] && NegQ[b^2-4*a*c]
```

8: $\int \frac{\sqrt{a+b x+c x^2}}{d+e x+f x^2} \, dx$ when $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b x+c x^2}}{d+e x+f x^2} == \frac{c}{f \sqrt{a+b x+c x^2}} - \frac{c d-a f+(c e-b f) x}{f \sqrt{a+b x+c x^2} \left(d+e x+f x^2\right)}$

Rule 1.2.1.5.8: If $b^2-4 a c \neq 0 \land e^2-4 d f \neq 0$, then

$$\int \frac{\sqrt{a+b x+c x^2}}{d+e x+f x^2} \, dx \rightarrow \frac{c}{f} \int \frac{1}{\sqrt{a+b x+c x^2}} \, dx - \frac{1}{f} \int \frac{c d-a f+(c e-b f) x}{\sqrt{a+b x+c x^2} \left(d+e x+f x^2\right)} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_+c_.*x_^2]/(d_+e_.*x_+f_.*x_^2),x_Symbol] :=
  c/f*Int[1/Sqrt[a+b*x+c*x^2],x] -
  1/f*Int[(c*d-a*f+(c*e-b*f)*x)/(Sqrt[a+b*x+c*x^2]*(d+e*x+f*x^2)),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0]
```

```
Int[Sqrt[a_+b_.*x_+c_.*x_^2]/(d_+f_.*x_^2),x_Symbol] :=
  c/f*Int[1/Sqrt[a+b*x+c*x^2],x] -
  1/f*Int[(c*d-a*f-b*f*x)/(Sqrt[a+b*x+c*x^2]*(d+f*x^2)),x] /;
FreeQ[{a,b,c,d,f},x] && NeQ[b^2-4*a*c,0]
```

```
Int[Sqrt[a_+c_.*x_^2]/(d_+e_.*x_+f_.*x_^2),x_Symbol] :=
  c/f*Int[1/Sqrt[a+c*x^2],x] -
  1/f*Int[(c*d-a*f+c*e*x)/(Sqrt[a+c*x^2]*(d+e*x+f*x^2)),x] /;
FreeQ[{a,c,d,e,f},x] && NeQ[e^2-4*d*f,0]
```

9: $\int \frac{1}{\sqrt{a+b x+c x^2}\,\sqrt{d+e x+f x^2}}\,dx$ when $b^2-4 a c\neq 0\wedge e^2-4 d f\neq 0$

Derivation: Piecewise constant extraction

Basis: Let $r\to\sqrt{b^2-4 a c}$, then $\partial_x \frac{\sqrt{b+r+2 c x}\,\sqrt{2 a+(b+r) x}}{\sqrt{a+b x+c x^2}}==0$

- Rule 1.2.1.5.9: If $b^2-4 a c\neq 0\wedge e^2-4 d f\neq 0$, let $r\to\sqrt{b^2-4 a c}$, then

$$\int \frac{1}{\sqrt{a+b x+c x^2}\,\sqrt{d+e x+f x^2}}\,dx\to\frac{\sqrt{b+r+2 c x}\,\sqrt{2 a+(b+r) x}}{\sqrt{a+b x+c x^2}}\int \frac{1}{\sqrt{b+r+2 c x}\,\sqrt{2 a+(b+r) x}\,\sqrt{d+e x+f x^2}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_+c_.*x_^2]*Sqrt[d_+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{r=Rt[b^2-4*a*c,2]},
  Sqrt[b+r+2*c*x]*Sqrt[2*a+(b+r)*x]/Sqrt[a+b*x+c*x^2]*Int[1/(Sqrt[b+r+2*c*x]*Sqrt[2*a+(b+r)*x]*Sqrt[d+e*x+f*x^2]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && NeQ[e^2-4*d*f,0]
```

```
Int[1/(Sqrt[a_+b_.*x_+c_.*x_^2]*Sqrt[d_+f_.*x_^2]),x_Symbol] :=
  With[{r=Rt[b^2-4*a*c,2]},
  Sqrt[b+r+2*c*x]*Sqrt[2*a+(b+r)*x]/Sqrt[a+b*x+c*x^2]*Int[1/(Sqrt[b+r+2*c*x]*Sqrt[2*a+(b+r)*x]*Sqrt[d+f*x^2]),x]] /;
FreeQ[{a,b,c,d,f},x] && NeQ[b^2-4*a*c,0]
```

X: $\int (a+b\,x+c\,x^2)^p\,(d+e\,x+f\,x^2)^q\,dx$

Rule 1.2.1.5.X:

$$\int (a+b\,x+c\,x^2)^p\,(d+e\,x+f\,x^2)^q\,dx \longrightarrow \int (a+b\,x+c\,x^2)^p\,(d+e\,x+f\,x^2)^q\,dx$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  Unintegrable[(a+b*x+c*x^2)^p*(d+e*x+f*x^2)^q,x] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && Not[IGtQ[p,0]] && Not[IGtQ[q,0]]
```

```
Int[(a_+c_.*x_^2)^p_*(d_.+e_.*x_+f_.*x_^2)^q_,x_Symbol] :=
  Unintegrable[(a+c*x^2)^p*(d+e*x+f*x^2)^q,x] /;
FreeQ[{a,c,d,e,f,p,q},x] && Not[IGtQ[p,0]] && Not[IGtQ[q,0]]
```

S: $\int (a + b\,u + c\,u^2)^p\,(d + e\,u + f\,u^2)^q\,dx$ when $u == g + h\,x$

Derivation: Integration by substitution

Rule 1.2.1.5.S: If $u == g + h\,x$, then

$$\int (a + b\,u + c\,u^2)^p\,(d + e\,u + f\,u^2)^q\,dx \longrightarrow \frac{1}{h}\,\text{Subst}\left[\int (a + b\,x + c\,x^2)^p\,(d + e\,x + f\,x^2)^q\,dx,\,x,\,u\right]$$

Program code:

```
Int[(a_.+b_.*u_+c_.*u_^2)^p_.*(d_.+e_.*u_+f_.*u_^2)^q_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x+c*x^2)^p*(d+e*x+f*x^2)^q,x],x,u] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(a_.+c_.*u_^2)^p_.*(d_.+e_.*u_+f_.*u_^2)^q_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+c*x^2)^p*(d+e*x+f*x^2)^q,x],x,u] /;
FreeQ[{a,c,d,e,f,p,q},x] && LinearQ[u,x] && NeQ[u,x]
```