Rules for integrands involving inverse sines and cosines

1. $\int u\,(a+b\,\mathrm{ArcSin}[c+d\,x])^n\,dx$

   1: $\int (a+b\,\mathrm{ArcSin}[c+d\,x])^n\,dx$

Derivation: Integration by substitution

Rule:

$$\int (a+b\,\mathrm{ArcSin}[c+d\,x])^n\,dx \to \frac{1}{d}\,\mathrm{Subst}\left[\int (a+b\,\mathrm{ArcSin}[x])^n\,dx,\ x,\ c+d\,x\right]$$

Program code:

```
Int[(a_.+b_.*ArcSin[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcSin[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[(a_.+b_.*ArcCos[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcCos[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int (e+f\,x)^m\,(a+b\,\text{ArcSin}[c+d\,x])^n\,dx$

Derivation: Integration by substitution

Rule:

$$\int (e+f\,x)^m\,(a+b\,\text{ArcSin}[c+d\,x])^n\,dx\;\longrightarrow\;\frac{1}{d}\,\text{Subst}\left[\int\left(\frac{d\,e-c\,f}{d}+\frac{f\,x}{d}\right)^m\,(a+b\,\text{ArcSin}[x])^n\,dx,\;x,\;c+d\,x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSin[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcSin[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCos[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcCos[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

3: $\int (A + B x + C x^2)^p (a + b \operatorname{ArcSin}[c + d x])^n \, dx$ when $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$

Derivation: Integration by substitution

Basis: If $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$, then $A + B x + C x^2 == -\frac{C}{d^2} + \frac{C}{d^2} (c + d x)^2$

Rule: If $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$, then

$$\int (A + B x + C x^2)^p (a + b \operatorname{ArcSin}[c + d x])^n \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int \left(-\frac{C}{d^2} + \frac{C x^2}{d^2}\right)^p (a + b \operatorname{ArcSin}[x])^n \, dx,\ x,\ c + d x\right]$$

Program code:

```
Int[(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(-C/d^2+C/d^2*x^2)^p*(a+b*ArcSin[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,A,B,C,n,p},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

```
Int[(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(-C/d^2+C/d^2*x^2)^p*(a+b*ArcCos[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,A,B,C,n,p},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

4: $\int (e + f x)^m \left(A + B x + C x^2\right)^p (a + b \operatorname{ArcSin}[c + d x])^n \, dx$ when $B \left(1 - c^2\right) + 2 A c d == 0 \land 2 c C - B d == 0$

Derivation: Integration by substitution

Basis: If $B \left(1 - c^2\right) + 2 A c d == 0 \land 2 c C - B d == 0$, then $A + B x + C x^2 == -\frac{C}{d^2} + \frac{C}{d^2} (c + d x)^2$

Rule: If $B \left(1 - c^2\right) + 2 A c d == 0 \land 2 c C - B d == 0$, then

$$\int (e + f x)^m \left(A + B x + C x^2\right)^p (a + b \operatorname{ArcSin}[c + d x])^n \, dx \rightarrow \frac{1}{d} \operatorname{Subst}\left[\int \left(\frac{d e - c f}{d} + \frac{f x}{d}\right)^m \left(-\frac{C}{d^2} + \frac{C x^2}{d^2}\right)^p (a + b \operatorname{ArcSin}[x])^n \, dx, x, c + d x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(-C/d^2+C/d^2*x^2)^p*(a+b*ArcSin[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n,p},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

```
Int[(e_.+f_.*x_)^m_.*(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(-C/d^2+C/d^2*x^2)^p*(a+b*ArcCos[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n,p},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

2. $\int \left(a + b \operatorname{ArcSin}\left[c + d x^2\right]\right)^n \, dx$ when $c^2 == 1$

1. $\int \left(a + b \operatorname{ArcSin}\left[c + d x^2\right]\right)^n \, dx$ when $c^2 == 1 \land n > 0$

1. $\int \sqrt{a + b \operatorname{ArcSin}\left[c + d x^2\right]} \, dx$ when $c^2 == 1$

1: $\int \sqrt{a + b \operatorname{ArcSin}\left[c + d x^2\right]} \, dx$ when $c^2 == 1$

Derivation: Integration by parts

Rule: If $c^2 == 1$, then

$$\int\sqrt{a+b\,\text{ArcSin}[c+d\,x^2]}\,dx\;\to\;x\sqrt{a+b\,\text{ArcSin}[c+d\,x^2]}-b\,d\int\frac{x^2}{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}\;\sqrt{a+b\,\text{ArcSin}[c+d\,x^2]}}\,dx$$

$$\to\;x\sqrt{a+b\,\text{ArcSin}[c+d\,x^2]}-$$

$$\frac{\sqrt{\pi}\;x\left(\text{Cos}\left[\frac{a}{2\,b}\right]+c\,\text{Sin}\left[\frac{a}{2\,b}\right]\right)\text{FresnelC}\left[\sqrt{\frac{c}{\pi\,b}}\;\sqrt{a+b\,\text{ArcSin}[c+d\,x^2]}\right]}{\sqrt{\frac{c}{b}}\;\left(\text{Cos}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]-c\,\text{Sin}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]\right)}+\frac{\sqrt{\pi}\;x\left(\text{Cos}\left[\frac{a}{2\,b}\right]-c\,\text{Sin}\left[\frac{a}{2\,b}\right]\right)\text{FresnelS}\left[\sqrt{\frac{c}{\pi\,b}}\;\sqrt{a+b\,\text{ArcSin}[c+d\,x^2]}\right]}{\sqrt{\frac{c}{b}}\;\left(\text{Cos}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]-c\,\text{Sin}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]\right)}$$

Program code:

```
Int[Sqrt[a_.+b_.*ArcSin[c_+d_.*x_^2]],x_Symbol] :=
  x*Sqrt[a+b*ArcSin[c+d*x^2]] -
  Sqrt[Pi]*x*(Cos[a/(2*b)]+c*Sin[a/(2*b)])*FresnelC[Sqrt[c/(Pi*b)]*Sqrt[a+b*ArcSin[c+d*x^2]]]/
     (Sqrt[c/b]*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) +
   Sqrt[Pi]*x*(Cos[a/(2*b)]-c*Sin[a/(2*b)])*FresnelS[Sqrt[c/(Pi*b)]*Sqrt[a+b*ArcSin[c+d*x^2]]]/
     (Sqrt[c/b]*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1]
```

2. $\int\sqrt{a+b\,\text{ArcCos}[c+d\,x^2]}\,dx$ when $c^2==1$

1: $\int\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}\,dx$

Rule:

$$\int\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}\,dx\;\to$$

$$-\frac{2\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}\;\text{Sin}\left[\frac{1}{2}\,\text{ArcCos}[1+d\,x^2]\right]^2}{d\,x}+$$

$$-\frac{1}{\sqrt{\frac{1}{b}}\;d\,x}2\sqrt{\pi}\;\text{Sin}\left[\frac{a}{2\,b}\right]\text{Sin}\left[\frac{1}{2}\,\text{ArcCos}[1+d\,x^2]\right]\text{FresnelC}\left[\sqrt{\frac{1}{\pi\,b}}\;\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}\right]+$$

$$\frac{1}{\sqrt{\frac{1}{b}}\ d\ x} 2\sqrt{\pi}\ \text{Cos}\left[\frac{a}{2\ b}\right]\ \text{Sin}\left[\frac{1}{2}\ \text{ArcCos}\left[1+d\ x^2\right]\right]\ \text{FresnelS}\left[\sqrt{\frac{1}{\pi\ b}}\ \sqrt{a+b\ \text{ArcCos}\left[1+d\ x^2\right]}\ \right]$$

Program code:

```
Int[Sqrt[a_.+b_.*ArcCos[1+d_.*x_^2]],x_Symbol] :=
  -2*Sqrt[a+b*ArcCos[1+d*x^2]]*Sin[ArcCos[1+d*x^2]/2]^2/(d*x) -
  2*Sqrt[Pi]*Sin[a/(2*b)]*Sin[ArcCos[1+d*x^2]/2]*FresnelC[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[1+d*x^2]]]/(Sqrt[1/b]*d*x) +
  2*Sqrt[Pi]*Cos[a/(2*b)]*Sin[ArcCos[1+d*x^2]/2]*FresnelS[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[1+d*x^2]]]/(Sqrt[1/b]*d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}\, dx$

Rule:

$$\int \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}\, dx \to$$
$$\frac{2 \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}\ \operatorname{Cos}\left[\frac{1}{2} \operatorname{ArcCos}[-1 + d x^2]\right]^2}{d x} -$$
$$\frac{1}{\sqrt{\frac{1}{b}}\ d x} 2 \sqrt{\pi}\ \operatorname{Cos}\left[\frac{a}{2 b}\right] \operatorname{Cos}\left[\frac{1}{2} \operatorname{ArcCos}[-1 + d x^2]\right] \operatorname{FresnelC}\left[\sqrt{\frac{1}{\pi b}}\ \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}\right] -$$
$$\frac{1}{\sqrt{\frac{1}{b}}\ d x} 2 \sqrt{\pi}\ \operatorname{Sin}\left[\frac{a}{2 b}\right] \operatorname{Cos}\left[\frac{1}{2} \operatorname{ArcCos}[-1 + d x^2]\right] \operatorname{FresnelS}\left[\sqrt{\frac{1}{\pi b}}\ \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}\right]$$

Program code:

```
Int[Sqrt[a_.+b_.*ArcCos[-1+d_.*x_^2]],x_Symbol] :=
  2*Sqrt[a+b*ArcCos[-1+d*x^2]]*Cos[(1/2)*ArcCos[-1+d*x^2]]^2/(d*x) -
  2*Sqrt[Pi]*Cos[a/(2*b)]*Cos[ArcCos[-1+d*x^2]/2]*FresnelC[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[-1+d*x^2]]]/(Sqrt[1/b]*d*x) -
  2*Sqrt[Pi]*Sin[a/(2*b)]*Cos[ArcCos[-1+d*x^2]/2]*FresnelS[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[-1+d*x^2]]]/(Sqrt[1/b]*d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int (a + b \operatorname{ArcSin}[c + d x^2])^n\, dx$ when $c^2 == 1 \land n > 1$

Derivation: Integration by parts twice

Basis: If $c^2 == 1$, then $\partial_x (a + b \operatorname{ArcSin}[c + d x^2])^n == \frac{2 b d n x (a+b \operatorname{ArcSin}[c+d x^2])^{n-1}}{\sqrt{-2 c d x^2 - d^2 x^4}}$

Basis: $\frac{x^2}{\sqrt{-d\,x^2\,(2\,c+d\,x^2)}} == -\partial_x \frac{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}}{d^2\,x}$

Rule: If $c^2 == 1 \land n > 1$, then

$$\int (a+b\,\text{ArcSin}[c+d\,x^2])^n\,dx \to x\,(a+b\,\text{ArcSin}[c+d\,x^2])^n - 2\,b\,d\,n\int \frac{x^2\,(a+b\,\text{ArcSin}[c+d\,x^2])^{n-1}}{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}}\,dx$$

$$\to x\,(a+b\,\text{ArcSin}[c+d\,x^2])^n + \frac{2\,b\,n\,\sqrt{-2\,c\,d\,x^2-d^2\,x^4}\,(a+b\,\text{ArcSin}[c+d\,x^2])^{n-1}}{d\,x} - 4\,b^2\,n\,(n-1)\int (a+b\,\text{ArcSin}[c+d\,x^2])^{n-2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcSin[c_+d_.*x_^2])^n_,x_Symbol] :=
  x*(a+b*ArcSin[c+d*x^2])^n +
  2*b*n*Sqrt[-2*c*d*x^2-d^2*x^4]*(a+b*ArcSin[c+d*x^2])^(n-1)/(d*x) -
  4*b^2*n*(n-1)*Int[(a+b*ArcSin[c+d*x^2])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1] && GtQ[n,1]
```

```
Int[(a_.+b_.*ArcCos[c_+d_.*x_^2])^n_,x_Symbol] :=
  x*(a+b*ArcCos[c+d*x^2])^n -
  2*b*n*Sqrt[-2*c*d*x^2-d^2*x^4]*(a+b*ArcCos[c+d*x^2])^(n-1)/(d*x) -
  4*b^2*n*(n-1)*Int[(a+b*ArcCos[c+d*x^2])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1] && GtQ[n,1]
```

2. $\int (a+b\,\text{ArcSin}[c+d\,x^2])^n\,dx$ when $c^2 == 1 \land n < 0$

1. $\int \frac{1}{a+b\,\text{ArcSin}[c+d\,x^2]}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{a+b\,\text{ArcSin}[c+d\,x^2]}\,dx$ when $c^2 == 1$

Rule: If $c^2 == 1$, then

$$\int \frac{1}{a + b\,\text{ArcSin}[c + d\,x^2]}\,dx \longrightarrow$$
$$-\frac{x\left(c\,\text{Cos}\left[\frac{a}{2b}\right] - \text{Sin}\left[\frac{a}{2b}\right]\right)\text{CosIntegral}\left[\frac{c}{2b}\left(a + b\,\text{ArcSin}[c + d\,x^2]\right)\right]}{2b\left(\text{Cos}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right] - c\,\text{Sin}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right]\right)} - \frac{x\left(c\,\text{Cos}\left[\frac{a}{2b}\right] + \text{Sin}\left[\frac{a}{2b}\right]\right)\text{SinIntegral}\left[\frac{c}{2b}\left(a + b\,\text{ArcSin}[c + d\,x^2]\right)\right]}{2b\left(\text{Cos}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right] - c\,\text{Sin}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right]\right)}$$

Program code:

```
Int[1/(a_.+b_.*ArcSin[c_+d_.*x_^2]),x_Symbol] :=
  -x*(c*Cos[a/(2*b)]-Sin[a/(2*b)])*CosIntegral[(c/(2*b))*(a+b*ArcSin[c+d*x^2])]/
    (2*b*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) -
  x*(c*Cos[a/(2*b)]+Sin[a/(2*b)])*SinIntegral[(c/(2*b))*(a+b*ArcSin[c+d*x^2])]/
    (2*b*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1]
```

2. $\int \frac{1}{a + b\,\text{ArcCos}[c + d\,x^2]}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{a + b\,\text{ArcCos}[1 + d\,x^2]}\,dx$

Rule:

$$\int \frac{1}{a + b\,\text{ArcCos}[1 + d\,x^2]}\,dx \longrightarrow$$
$$\frac{x\,\text{Cos}\left[\frac{a}{2b}\right]\text{CosIntegral}\left[\frac{1}{2b}\left(a + b\,\text{ArcCos}[1 + d\,x^2]\right)\right]}{\sqrt{2}\,b\,\sqrt{-d\,x^2}} + \frac{x\,\text{Sin}\left[\frac{a}{2b}\right]\text{SinIntegral}\left[\frac{1}{2b}\left(a + b\,\text{ArcCos}[1 + d\,x^2]\right)\right]}{\sqrt{2}\,b\,\sqrt{-d\,x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCos[1+d_.*x_^2]),x_Symbol] :=
  x*Cos[a/(2*b)]*CosIntegral[(a+b*ArcCos[1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[-d*x^2]) +
  x*Sin[a/(2*b)]*SinIntegral[(a+b*ArcCos[1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[-d*x^2]) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{a + b \operatorname{ArcCos}[-1 + d x^2]} \, dx$

Rule:

$$\int \frac{1}{a + b \operatorname{ArcCos}[-1 + d x^2]} \, dx \longrightarrow$$
$$\frac{x \operatorname{Sin}\left[\frac{a}{2b}\right] \operatorname{CosIntegral}\left[\frac{1}{2b}\left(a + b \operatorname{ArcCos}[-1 + d x^2]\right)\right]}{\sqrt{2}\, b \sqrt{d x^2}} - \frac{x \operatorname{Cos}\left[\frac{a}{2b}\right] \operatorname{SinIntegral}\left[\frac{1}{2b}\left(a + b \operatorname{ArcCos}[-1 + d x^2]\right)\right]}{\sqrt{2}\, b \sqrt{d x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCos[-1+d_.*x_^2]),x_Symbol] :=
  x*Sin[a/(2*b)]*CosIntegral[(a+b*ArcCos[-1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[d*x^2]) -
  x*Cos[a/(2*b)]*SinIntegral[(a+b*ArcCos[-1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[d*x^2]) /;
FreeQ[{a,b,d},x]
```

2. $\int \frac{1}{\sqrt{a + b\,\text{ArcSin}[c + d\,x^2]}}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{\sqrt{a + b\,\text{ArcSin}[c + d\,x^2]}}\,dx$ when $c^2 == 1$

Rule: If $c^2 == 1$, then

$$\int \frac{1}{\sqrt{a + b\,\text{ArcSin}[c + d\,x^2]}}\,dx \longrightarrow$$

$$-\left(\left(\sqrt{\pi}\;x\left(\text{Cos}\left[\frac{a}{2\,b}\right] - c\,\text{Sin}\left[\frac{a}{2\,b}\right]\right)\text{FresnelC}\left[\frac{1}{\sqrt{b\,c}\,\sqrt{\pi}}\sqrt{a + b\,\text{ArcSin}[c + d\,x^2]}\right]\right) \Big/ \left(\sqrt{b\,c}\left(\text{Cos}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right] - c\,\text{Sin}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right]\right)\right)\right) -$$
$$\left(\sqrt{\pi}\;x\left(\text{Cos}\left[\frac{a}{2\,b}\right] + c\,\text{Sin}\left[\frac{a}{2\,b}\right]\right)\text{FresnelS}\left[\frac{1}{\sqrt{b\,c}\,\sqrt{\pi}}\sqrt{a + b\,\text{ArcSin}[c + d\,x^2]}\right]\right) \Big/ \left(\sqrt{b\,c}\left(\text{Cos}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right] - c\,\text{Sin}\left[\frac{1}{2}\text{ArcSin}[c + d\,x^2]\right]\right)\right)$$

Program code:

```
Int[1/Sqrt[a_.+b_.*ArcSin[c_+d_.*x_^2]],x_Symbol] :=
  -Sqrt[Pi]*x*(Cos[a/(2*b)]-c*Sin[a/(2*b)])*FresnelC[1/(Sqrt[b*c]*Sqrt[Pi])*Sqrt[a+b*ArcSin[c+d*x^2]]]/
    (Sqrt[b*c]*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) -
  Sqrt[Pi]*x*(Cos[a/(2*b)]+c*Sin[a/(2*b)])*FresnelS[(1/(Sqrt[b*c]*Sqrt[Pi]))*Sqrt[a+b*ArcSin[c+d*x^2]]]/
    (Sqrt[b*c]*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1]
```

2. $\int \frac{1}{\sqrt{a + b\,\text{ArcCos}[c + d\,x^2]}}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{\sqrt{a + b\,\text{ArcCos}[1 + d\,x^2]}}\,dx$

Rule:

$$\int \frac{1}{\sqrt{a+b\,\text{ArcCos}\left[1+d\,x^2\right]}}\,dx \to$$
$$-\frac{1}{d\,x}2\sqrt{\frac{\pi}{b}}\,\text{Cos}\left[\frac{a}{2\,b}\right]\,\text{Sin}\left[\frac{1}{2}\,\text{ArcCos}\left[1+d\,x^2\right]\right]\,\text{FresnelC}\left[\sqrt{\frac{1}{\pi\,b}}\,\sqrt{a+b\,\text{ArcCos}\left[1+d\,x^2\right]}\right]-$$
$$\frac{1}{d\,x}2\sqrt{\frac{\pi}{b}}\,\text{Sin}\left[\frac{a}{2\,b}\right]\,\text{Sin}\left[\frac{1}{2}\,\text{ArcCos}\left[1+d\,x^2\right]\right]\,\text{FresnelS}\left[\sqrt{\frac{1}{\pi\,b}}\,\sqrt{a+b\,\text{ArcCos}\left[1+d\,x^2\right]}\right]$$

Program code:

```
Int[1/Sqrt[a_.+b_.*ArcCos[1+d_.*x_^2]],x_Symbol] :=
  -2*Sqrt[Pi/b]*Cos[a/(2*b)]*Sin[ArcCos[1+d*x^2]/2]*FresnelC[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[1+d*x^2]]]/(d*x) -
   2*Sqrt[Pi/b]*Sin[a/(2*b)]*Sin[ArcCos[1+d*x^2]/2]*FresnelS[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[1+d*x^2]]]/(d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{\sqrt{a+b\,\text{ArcCos}\left[-1+d\,x^2\right]}}\,dx$

Rule:

$$\int \frac{1}{\sqrt{a+b\,\text{ArcCos}\left[-1+d\,x^2\right]}}\,dx \to$$
$$\frac{1}{d\,x}2\sqrt{\frac{\pi}{b}}\,\text{Sin}\left[\frac{a}{2\,b}\right]\,\text{Cos}\left[\frac{1}{2}\,\text{ArcCos}\left[-1+d\,x^2\right]\right]\,\text{FresnelC}\left[\sqrt{\frac{1}{\pi\,b}}\,\sqrt{a+b\,\text{ArcCos}\left[-1+d\,x^2\right]}\right]-$$
$$\frac{1}{d\,x}2\sqrt{\frac{\pi}{b}}\,\text{Cos}\left[\frac{a}{2\,b}\right]\,\text{Cos}\left[\frac{1}{2}\,\text{ArcCos}\left[-1+d\,x^2\right]\right]\,\text{FresnelS}\left[\sqrt{\frac{1}{\pi\,b}}\,\sqrt{a+b\,\text{ArcCos}\left[-1+d\,x^2\right]}\right]$$

Program code:

```
Int[1/Sqrt[a_.+b_.*ArcCos[-1+d_.*x_^2]],x_Symbol] :=
  2*Sqrt[Pi/b]*Sin[a/(2*b)]*Cos[ArcCos[-1+d*x^2]/2]*FresnelC[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[-1+d*x^2]]]/(d*x) -
  2*Sqrt[Pi/b]*Cos[a/(2*b)]*Cos[ArcCos[-1+d*x^2]/2]*FresnelS[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[-1+d*x^2]]]/(d*x) /;
FreeQ[{a,b,d},x]
```

3. $\int (a + b \operatorname{ArcSin}[c + d x^2])^n \, dx$ when $c^2 == 1 \land n < -1$

   1. $\int \frac{1}{(a + b \operatorname{ArcSin}[c + d x^2])^{3/2}} \, dx$ when $c^2 == 1$

      1: $\int \frac{1}{(a + b \operatorname{ArcSin}[c + d x^2])^{3/2}} \, dx$ when $c^2 == 1$

Derivation: Integration by parts

Basis: If $c^2 == 1$, then $-\frac{b\,d\,x}{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}\,(a+b \operatorname{ArcSin}[c+d\,x^2])^{3/2}} == \partial_x \frac{1}{\sqrt{a+b \operatorname{ArcSin}[c+d\,x^2]}}$

Rule: If $c^2 == 1$, then

$$\int \frac{1}{(a + b \operatorname{ArcSin}[c + d x^2])^{3/2}} \, dx \rightarrow -\frac{\sqrt{-2\,c\,d\,x^2 - d^2\,x^4}}{b\,d\,x\,\sqrt{a + b \operatorname{ArcSin}[c + d\,x^2]}} - \frac{d}{b} \int \frac{x^2}{\sqrt{-2\,c\,d\,x^2 - d^2\,x^4}\,\sqrt{a + b \operatorname{ArcSin}[c + d\,x^2]}} \, dx$$

$$\rightarrow -\frac{\sqrt{-2\,c\,d\,x^2 - d^2\,x^4}}{b\,d\,x\,\sqrt{a + b \operatorname{ArcSin}[c + d\,x^2]}} -$$

$$\left(\left(\frac{c}{b}\right)^{3/2} \sqrt{\pi}\, x \left(\operatorname{Cos}\left[\frac{a}{2\,b}\right] + c \operatorname{Sin}\left[\frac{a}{2\,b}\right]\right) \operatorname{FresnelC}\left[\sqrt{\frac{c}{\pi\,b}}\,\sqrt{a + b \operatorname{ArcSin}[c + d\,x^2]}\right]\right) \Big/ \left(\operatorname{Cos}\left[\frac{1}{2} \operatorname{ArcSin}[c + d\,x^2]\right] - c \operatorname{Sin}\left[\frac{1}{2} \operatorname{ArcSin}[c + d\,x^2]\right]\right) +$$

$$\left(\left(\frac{c}{b}\right)^{3/2} \sqrt{\pi}\, x \left(\operatorname{Cos}\left[\frac{a}{2\,b}\right] - c \operatorname{Sin}\left[\frac{a}{2\,b}\right]\right) \operatorname{FresnelS}\left[\sqrt{\frac{c}{\pi\,b}}\,\sqrt{a + b \operatorname{ArcSin}[c + d\,x^2]}\right]\right) \Big/ \left(\operatorname{Cos}\left[\frac{1}{2} \operatorname{ArcSin}[c + d\,x^2]\right] - c \operatorname{Sin}\left[\frac{1}{2} \operatorname{ArcSin}[c + d\,x^2]\right]\right)$$

Program code:

```
Int[1/(a_.+b_.*ArcSin[c_+d_.*x_^2])^(3/2),x_Symbol] :=
  -Sqrt[-2*c*d*x^2-d^2*x^4]/(b*d*x*Sqrt[a+b*ArcSin[c+d*x^2]]) -
  (c/b)^(3/2)*Sqrt[Pi]*x*(Cos[a/(2*b)]+c*Sin[a/(2*b)])*FresnelC[Sqrt[c/(Pi*b)]*Sqrt[a+b*ArcSin[c+d*x^2]]]/
    (Cos[(1/2)*ArcSin[c+d*x^2]]-c*Sin[ArcSin[c+d*x^2]/2]) +
  (c/b)^(3/2)*Sqrt[Pi]*x*(Cos[a/(2*b)]-c*Sin[a/(2*b)])*FresnelS[Sqrt[c/(Pi*b)]*Sqrt[a+b*ArcSin[c+d*x^2]]]/
    (Cos[(1/2)*ArcSin[c+d*x^2]]-c*Sin[ArcSin[c+d*x^2]/2]) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1]
```

2. $\int \frac{1}{(a+b\,\text{ArcCos}[c+d\,x^2])^{3/2}}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{(a+b\,\text{ArcCos}[1+d\,x^2])^{3/2}}\,dx$

Derivation: Integration by parts

Basis: $\frac{b\,d\,x}{\sqrt{-2\,d\,x^2-d^2\,x^4}\,(a+b\,\text{ArcCos}[1+d\,x^2])^{3/2}} == \partial_x \frac{1}{\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}}$

Rule:

$$\int \frac{1}{(a+b\,\text{ArcCos}[1+d\,x^2])^{3/2}}\,dx \to \frac{\sqrt{-2\,d\,x^2-d^2\,x^4}}{b\,d\,x\,\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}} + \frac{d}{b}\int \frac{x^2}{\sqrt{-2\,d\,x^2-d^2\,x^4}\,\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}}\,dx$$

$$\to \frac{\sqrt{-2\,d\,x^2-d^2\,x^4}}{b\,d\,x\,\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}} - \frac{1}{d\,x}2\left(\frac{1}{b}\right)^{3/2}\sqrt{\pi}\,\text{Sin}\left[\frac{a}{2\,b}\right]\,\text{Sin}\left[\frac{1}{2}\,\text{ArcCos}[1+d\,x^2]\right]\,\text{FresnelC}\left[\sqrt{\frac{1}{\pi\,b}}\,\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}\right] + \frac{1}{d\,x}2\left(\frac{1}{b}\right)^{3/2}\sqrt{\pi}\,\text{Cos}\left[\frac{a}{2\,b}\right]\,\text{Sin}\left[\frac{1}{2}\,\text{ArcCos}[1+d\,x^2]\right]\,\text{FresnelS}\left[\sqrt{\frac{1}{\pi\,b}}\,\sqrt{a+b\,\text{ArcCos}[1+d\,x^2]}\right]$$

Program code:

```
Int[1/(a_.+b_.*ArcCos[1+d_.*x_^2])^(3/2),x_Symbol] :=
  Sqrt[-2*d*x^2-d^2*x^4]/(b*d*x*Sqrt[a+b*ArcCos[1+d*x^2]]) -
  2*(1/b)^(3/2)*Sqrt[Pi]*Sin[a/(2*b)]*Sin[ArcCos[1+d*x^2]/2]*FresnelC[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[1+d*x^2]]]/(d*x) +
  2*(1/b)^(3/2)*Sqrt[Pi]*Cos[a/(2*b)]*Sin[ArcCos[1+d*x^2]/2]*FresnelS[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[1+d*x^2]]]/(d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{(a + b \operatorname{ArcCos}[-1 + d x^2])^{3/2}} \, dx$

Derivation: Integration by parts

Basis: $\frac{b d x}{\sqrt{2 d x^2 - d^2 x^4} \, (a + b \operatorname{ArcCos}[-1 + d x^2])^{3/2}} == \partial_x \frac{1}{\sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}}$

Rule:

$$\int \frac{1}{(a + b \operatorname{ArcCos}[-1 + d x^2])^{3/2}} \, dx \to \frac{\sqrt{2 d x^2 - d^2 x^4}}{b d x \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}} + \frac{d}{b} \int \frac{x^2}{\sqrt{2 d x^2 - d^2 x^4} \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}} \, dx$$

$$\to \frac{\sqrt{2 d x^2 - d^2 x^4}}{b d x \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}} - \frac{1}{d x} 2 \left(\frac{1}{b}\right)^{3/2} \sqrt{\pi} \operatorname{Cos}\left[\frac{a}{2 b}\right] \operatorname{Cos}\left[\frac{1}{2} \operatorname{ArcCos}[-1 + d x^2]\right] \operatorname{FresnelC}\left[\sqrt{\frac{1}{\pi b}} \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}\right] - \frac{1}{d x} 2 \left(\frac{1}{b}\right)^{3/2} \sqrt{\pi} \operatorname{Sin}\left[\frac{a}{2 b}\right] \operatorname{Cos}\left[\frac{1}{2} \operatorname{ArcCos}[-1 + d x^2]\right] \operatorname{FresnelS}\left[\sqrt{\frac{1}{\pi b}} \sqrt{a + b \operatorname{ArcCos}[-1 + d x^2]}\right]$$

Program code:

```
Int[1/(a_.+b_.*ArcCos[-1+d_.*x_^2])^(3/2),x_Symbol] :=
  Sqrt[2*d*x^2-d^2*x^4]/(b*d*x*Sqrt[a+b*ArcCos[-1+d*x^2]]) -
  2*(1/b)^(3/2)*Sqrt[Pi]*Cos[a/(2*b)]*Cos[ArcCos[-1+d*x^2]/2]*FresnelC[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[-1+d*x^2]]]/(d*x) -
  2*(1/b)^(3/2)*Sqrt[Pi]*Sin[a/(2*b)]*Cos[ArcCos[-1+d*x^2]/2]*FresnelS[Sqrt[1/(Pi*b)]*Sqrt[a+b*ArcCos[-1+d*x^2]]]/(d*x) /;
FreeQ[{a,b,d},x]
```

2. $\int \frac{1}{(a+b\,\text{ArcSin}[c+d\,x^2])^2}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{(a+b\,\text{ArcSin}[c+d\,x^2])^2}\,dx$ when $c^2 == 1$

Derivation: Integration by parts

Basis: If $c^2 == 1$, then $-\frac{2\,b\,d\,x}{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}\,(a+b\,\text{ArcSin}[c+d\,x^2])^2} == \partial_x \frac{1}{a+b\,\text{ArcSin}[c+d\,x^2]}$

Rule: If $c^2 == 1$, then

$$\int \frac{1}{(a+b\,\text{ArcSin}[c+d\,x^2])^2}\,dx \to -\frac{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}}{2\,b\,d\,x\,(a+b\,\text{ArcSin}[c+d\,x^2])} - \frac{d}{2\,b}\int \frac{x^2}{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}\,(a+b\,\text{ArcSin}[c+d\,x^2])}\,dx$$

$$\to -\frac{\sqrt{-2\,c\,d\,x^2-d^2\,x^4}}{2\,b\,d\,x\,(a+b\,\text{ArcSin}[c+d\,x^2])} - \frac{x\left(\text{Cos}\left[\frac{a}{2\,b}\right]+c\,\text{Sin}\left[\frac{a}{2\,b}\right]\right)\text{CosIntegral}\left[\frac{c}{2\,b}\,(a+b\,\text{ArcSin}[c+d\,x^2])\right]}{4\,b^2\left(\text{Cos}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]-c\,\text{Sin}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]\right)} + \frac{x\left(\text{Cos}\left[\frac{a}{2\,b}\right]-c\,\text{Sin}\left[\frac{a}{2\,b}\right]\right)\text{SinIntegral}\left[\frac{c}{2\,b}\,(a+b\,\text{ArcSin}[c+d\,x^2])\right]}{4\,b^2\left(\text{Cos}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]-c\,\text{Sin}\left[\frac{1}{2}\,\text{ArcSin}[c+d\,x^2]\right]\right)}$$

Program code:

```
Int[1/(a_.+b_.*ArcSin[c_+d_.*x_^2])^2,x_Symbol] :=
  -Sqrt[-2*c*d*x^2-d^2*x^4]/(2*b*d*x*(a+b*ArcSin[c+d*x^2])) -
  x*(Cos[a/(2*b)]+c*Sin[a/(2*b)])*CosIntegral[(c/(2*b))*(a+b*ArcSin[c+d*x^2])]/
    (4*b^2*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) +
  x*(Cos[a/(2*b)]-c*Sin[a/(2*b)])*SinIntegral[(c/(2*b))*(a+b*ArcSin[c+d*x^2])]/
    (4*b^2*(Cos[ArcSin[c+d*x^2]/2]-c*Sin[ArcSin[c+d*x^2]/2])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1]
```

2. $\int \frac{1}{(a + b\,\text{ArcCos}[c + d\,x^2])^2}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{(a + b\,\text{ArcCos}[1 + d\,x^2])^2}\,dx$

Rule:

$$\int \frac{1}{(a + b\,\text{ArcCos}[1 + d\,x^2])^2}\,dx \longrightarrow$$
$$\frac{\sqrt{-2\,d\,x^2 - d^2\,x^4}}{2\,b\,d\,x\,(a + b\,\text{ArcCos}[1 + d\,x^2])} + \frac{x\,\text{Sin}\left[\frac{a}{2\,b}\right]\,\text{CosIntegral}\left[\frac{1}{2\,b}\,(a + b\,\text{ArcCos}[1 + d\,x^2])\right]}{2\,\sqrt{2}\,b^2\,\sqrt{-d\,x^2}} - \frac{x\,\text{Cos}\left[\frac{a}{2\,b}\right]\,\text{SinIntegral}\left[\frac{1}{2\,b}\,(a + b\,\text{ArcCos}[1 + d\,x^2])\right]}{2\,\sqrt{2}\,b^2\,\sqrt{-d\,x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCos[1+d_.*x_^2])^2,x_Symbol] :=
  Sqrt[-2*d*x^2-d^2*x^4]/(2*b*d*x*(a+b*ArcCos[1+d*x^2])) +
  x*Sin[a/(2*b)]*CosIntegral[(a+b*ArcCos[1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[(-d)*x^2]) -
  x*Cos[a/(2*b)]*SinIntegral[(a+b*ArcCos[1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[(-d)*x^2]) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{(a + b\,\text{ArcCos}[-1 + d\,x^2])^2}\,dx$

Rule:

$$\int \frac{1}{(a + b\,\text{ArcCos}[-1 + d\,x^2])^2}\,dx \to$$
$$\frac{\sqrt{2\,d\,x^2 - d^2\,x^4}}{2\,b\,d\,x\,(a + b\,\text{ArcCos}[-1 + d\,x^2])} - \frac{x\,\text{Cos}\left[\frac{a}{2\,b}\right]\,\text{CosIntegral}\left[\frac{1}{2\,b}\,(a + b\,\text{ArcCos}[-1 + d\,x^2])\right]}{2\,\sqrt{2}\,b^2\,\sqrt{d\,x^2}} - \frac{x\,\text{Sin}\left[\frac{a}{2\,b}\right]\,\text{SinIntegral}\left[\frac{1}{2\,b}\,(a + b\,\text{ArcCos}[-1 + d\,x^2])\right]}{2\,\sqrt{2}\,b^2\,\sqrt{d\,x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCos[-1+d_.*x_^2])^2,x_Symbol] :=
  Sqrt[2*d*x^2-d^2*x^4]/(2*b*d*x*(a+b*ArcCos[-1+d*x^2])) -
  x*Cos[a/(2*b)]*CosIntegral[(a+b*ArcCos[-1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[d*x^2]) -
  x*Sin[a/(2*b)]*SinIntegral[(a+b*ArcCos[-1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[d*x^2]) /;
FreeQ[{a,b,d},x]
```

3: $\int \left(a + b\,\mathrm{ArcSin}\left[c + d\,x^2\right]\right)^n \,dx$ when $c^2 == 1 \land n < -1 \land n \neq -2$

Derivation: Inverted integration by parts twice

Rule: If $c^2 == 1 \land n < -1 \land n \neq -2$, then

$$\int \left(a + b\,\mathrm{ArcSin}\left[c + d\,x^2\right]\right)^n \,dx \longrightarrow$$

$$\frac{x\,\left(a + b\,\mathrm{ArcSin}\left[c + d\,x^2\right]\right)^{n+2}}{4\,b^2\,(n+1)\,(n+2)} + \frac{\sqrt{-2\,c\,d\,x^2 - d^2\,x^4}\,\left(a + b\,\mathrm{ArcSin}\left[c + d\,x^2\right]\right)^{n+1}}{2\,b\,d\,(n+1)\,x} - \frac{1}{4\,b^2\,(n+1)\,(n+2)} \int \left(a + b\,\mathrm{ArcSin}\left[c + d\,x^2\right]\right)^{n+2} \,dx$$

Program code:

```
Int[(a_.+b_.*ArcSin[c_+d_.*x_^2])^n_,x_Symbol] :=
  x*(a+b*ArcSin[c+d*x^2])^(n+2)/(4*b^2*(n+1)*(n+2)) +
  Sqrt[-2*c*d*x^2-d^2*x^4]*(a+b*ArcSin[c+d*x^2])^(n+1)/(2*b*d*(n+1)*x) -
  1/(4*b^2*(n+1)*(n+2))*Int[(a+b*ArcSin[c+d*x^2])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1] && LtQ[n,-1] && NeQ[n,-2]
```

```
Int[(a_.+b_.*ArcCos[c_+d_.*x_^2])^n_,x_Symbol] :=
  x*(a+b*ArcCos[c+d*x^2])^(n+2)/(4*b^2*(n+1)*(n+2)) -
  Sqrt[-2*c*d*x^2-d^2*x^4]*(a+b*ArcCos[c+d*x^2])^(n+1)/(2*b*d*(n+1)*x) -
  1/(4*b^2*(n+1)*(n+2))*Int[(a+b*ArcCos[c+d*x^2])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1] && LtQ[n,-1] && NeQ[n,-2]
```

3: $\int \frac{\text{ArcSin}[a\,x^p]^n}{x}\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $\frac{\text{ArcSin}[a\,x^p]^n}{x} = \frac{1}{p}\,\text{ArcSin}[a\,x^p]^n\,\text{Cot}[\text{ArcSin}[a\,x^p]]\,\partial_x\,\text{ArcSin}[a\,x^p]$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \frac{\text{ArcSin}[a\,x^p]^n}{x}\,dx \longrightarrow \frac{1}{p}\,\text{Subst}\left[\int x^n\,\text{Cot}[x]\,dx,\,x,\,\text{ArcSin}[a\,x^p]\right]$$

Program code:

```
Int[ArcSin[a_.*x_^p_]^n_./x_,x_Symbol] :=
  1/p*Subst[Int[x^n*Cot[x],x],x,ArcSin[a*x^p]] /;
FreeQ[{a,p},x] && IGtQ[n,0]
```

```
Int[ArcCos[a_.*x_^p_]^n_./x_,x_Symbol] :=
  -1/p*Subst[Int[x^n*Tan[x],x],x,ArcCos[a*x^p]] /;
FreeQ[{a,p},x] && IGtQ[n,0]
```

4: $\int u\,\text{ArcSin}\left[\frac{c}{a+b\,x^n}\right]^m dx$

Derivation: Algebraic simplification

Basis: $\text{ArcSin}[z] = \text{ArcCsc}\left[\frac{1}{z}\right]$

Rule:

$$\int u\,\text{ArcSin}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u\,\text{ArcCsc}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

Program code:

```
Int[u_.*ArcSin[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCsc[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

```
Int[u_.*ArcCos[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcSec[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

5: $\int \frac{\text{ArcSin}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{1+b x^2}} \, dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\sqrt{-b x^2}}{x} == 0$

Basis: $\frac{x \, \text{ArcSin}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{-b x^2} \sqrt{1+b x^2}} == \frac{1}{b} \, \text{Subst}\left[\frac{\text{ArcSin}[x]^n}{\sqrt{1-x^2}}, x, \sqrt{1+b x^2}\right] \partial_x \sqrt{1+b x^2}$

Rule:

$$\int \frac{\text{ArcSin}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{1+b x^2}} \, dx \to \frac{\sqrt{-b x^2}}{x} \int \frac{x \, \text{ArcSin}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{-b x^2} \sqrt{1+b x^2}} \, dx$$

$$\to \frac{\sqrt{-b x^2}}{b x} \, \text{Subst}\left[\int \frac{\text{ArcSin}[x]^n}{\sqrt{1-x^2}} \, dx, x, \sqrt{1+b x^2}\right]$$

Program code:

```
Int[ArcSin[Sqrt[1+b_.*x_^2]]^n_./Sqrt[1+b_.*x_^2],x_Symbol] :=
  Sqrt[-b*x^2]/(b*x)*Subst[Int[ArcSin[x]^n/Sqrt[1-x^2],x],x,Sqrt[1+b*x^2]] /;
FreeQ[{b,n},x]
```

```
Int[ArcCos[Sqrt[1+b_.*x_^2]]^n_./Sqrt[1+b_.*x_^2],x_Symbol] :=
  Sqrt[-b*x^2]/(b*x)*Subst[Int[ArcCos[x]^n/Sqrt[1-x^2],x],x,Sqrt[1+b*x^2]] /;
FreeQ[{b,n},x]
```

6: $\int u\, f^{c\,\text{ArcSin}[a+b\,x]^n}\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis:
$F[x, \text{ArcSin}[a+b\,x]] == \frac{1}{b}\,\text{Subst}\left[F\left[-\frac{a}{b}+\frac{\text{Sin}[x]}{b}, x\right]\text{Cos}[x], x, \text{ArcSin}[a+b\,x]\right]\partial_x\text{ArcSin}[a+b\,x]$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int u\, f^{c\,\text{ArcSin}[a+b\,x]^n}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int \text{Subst}\left[u, x, -\frac{a}{b}+\frac{\text{Sin}[x]}{b}\right] f^{c\,x^n}\,\text{Cos}[x]\,dx, x, \text{ArcSin}[a+b\,x]\right]$$

Program code:

```
Int[u_.*f_^(c_.*ArcSin[a_.+b_.*x_]^n_.),x_Symbol] :=
  1/b*Subst[Int[ReplaceAll[u,x→-a/b+Sin[x]/b]*f^(c*x^n)*Cos[x],x],x,ArcSin[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[n,0]
```

```
Int[u_.*f_^(c_.*ArcCos[a_.+b_.*x_]^n_.),x_Symbol] :=
  -1/b*Subst[Int[ReplaceAll[u,x→-a/b+Cos[x]/b]*f^(c*x^n)*Sin[x],x],x,ArcCos[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[n,0]
```

7. $\int v\,(a+b\,\text{ArcSin}[u])\,dx$ when **u is free of inverse functions**

   1. $\int v\,(a+b\,\text{ArcSin}[u])\,dx$ when **u is free of inverse functions**

      1: $\int \text{ArcSin}\left[a\,x^2+b\,\sqrt{c+d\,x^2}\right]dx$ when $b^2\,c == 1$

Derivation: Integration by parts and piecewise constant extraction

Basis: If $b^2\,c == 1$, then $1-\left(a\,x^2+b\,\sqrt{c+d\,x^2}\right)^2 == -x^2\left(b^2\,d+a^2\,x^2+2\,a\,b\,\sqrt{c+d\,x^2}\right)$

Basis: $\partial_x \frac{x \sqrt{b^2 d+a^2 x^2+2 a b \sqrt{c+d x^2}}}{\sqrt{-x^2 \left(b^2 d+a^2 x^2+2 a b \sqrt{c+d x^2}\right)}} == 0$

Note: The resulting integrand is of the form $x\ F\left[x^2\right]$ which can be integrated by substitution.

Rule: If $b^2\ c == 1$, then

$$\int \text{ArcSin}\left[a x^2+b \sqrt{c+d x^2}\right] dx \longrightarrow x\, \text{ArcSin}\left[a x^2+b \sqrt{c+d x^2}\right]-\int \frac{x^2 \left(b d+2 a \sqrt{c+d x^2}\right)}{\sqrt{c+d x^2} \sqrt{-x^2 \left(b^2 d+a^2 x^2+2 a b \sqrt{c+d x^2}\right)}} dx$$

$$\longrightarrow x\, \text{ArcSin}\left[a x^2+b \sqrt{c+d x^2}\right]-\frac{x \sqrt{b^2 d+a^2 x^2+2 a b \sqrt{c+d x^2}}}{\sqrt{-x^2 \left(b^2 d+a^2 x^2+2 a b \sqrt{c+d x^2}\right)}} \int \frac{x \left(b d+2 a \sqrt{c+d x^2}\right)}{\sqrt{c+d x^2} \sqrt{b^2 d+a^2 x^2+2 a b \sqrt{c+d x^2}}} dx$$

Program code:

```
Int[ArcSin[a_.*x_^2+b_.*Sqrt[c_+d_.*x_^2]],x_Symbol] :=
  x*ArcSin[a*x^2+b*Sqrt[c+d*x^2]] -
  x*Sqrt[b^2*d+a^2*x^2+2*a*b*Sqrt[c+d*x^2]]/Sqrt[(-x^2)*(b^2*d+a^2*x^2+2*a*b*Sqrt[c+d*x^2])]*
    Int[x*(b*d+2*a*Sqrt[c+d*x^2])/(Sqrt[c+d*x^2]*Sqrt[b^2*d +a^2*x^2+2*a*b*Sqrt[c+d*x^2]]),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2*c,1]
```

```
Int[ArcCos[a_.*x_^2+b_.*Sqrt[c_+d_.*x_^2]],x_Symbol] :=
  x*ArcCos[a*x^2+b*Sqrt[c+d*x^2]] +
  x*Sqrt[b^2*d+a^2*x^2+2*a*b*Sqrt[c+d*x^2]]/Sqrt[(-x^2)*(b^2*d+a^2*x^2+2*a*b*Sqrt[c+d*x^2])]*
    Int[x*(b*d+2*a*Sqrt[c+d*x^2])/(Sqrt[c+d*x^2]*Sqrt[b^2*d+a^2*x^2+2*a*b*Sqrt[c+d*x^2]]),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2*c,1]
```

2: $\int \text{ArcSin}[u] \, dx$ when u is free of inverse functions

Derivation: Integration by parts

Rule: If u is free of inverse functions, then

$$\int \text{ArcSin}[u] \, dx \longrightarrow x \, \text{ArcSin}[u] - \int \frac{x \, \partial_x u}{\sqrt{1-u^2}} \, dx$$

Program code:

```
Int[ArcSin[u_],x_Symbol] :=
  x*ArcSin[u] -
  Int[SimplifyIntegrand[x*D[u,x]/Sqrt[1-u^2],x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

```
Int[ArcCos[u_],x_Symbol] :=
  x*ArcCos[u] +
  Int[SimplifyIntegrand[x*D[u,x]/Sqrt[1-u^2],x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

2: $\int (c + d\,x)^m\,(a + b\,\text{ArcSin}[u])\,dx$ when $m \neq -1 \land u$ is free of inverse functions

Derivation: Integration by parts

Rule: If $m \neq -1 \land u$ is free of inverse functions, then

$$\int (c + d\,x)^m\,(a + b\,\text{ArcSin}[u])\,dx \longrightarrow \frac{(c + d\,x)^{m+1}\,(a + b\,\text{ArcSin}[u])}{d\,(m + 1)} - \frac{b}{d\,(m + 1)} \int \frac{(c + d\,x)^{m+1}\,\partial_x u}{\sqrt{1 - u^2}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcSin[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcSin[u])/(d*(m+1)) -
  b/(d*(m+1))*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/Sqrt[1-u^2],x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponentialQ[u,
```

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcCos[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcCos[u])/(d*(m+1)) +
  b/(d*(m+1))*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/Sqrt[1-u^2],x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponentialQ[u,
```

3: $\int v\,(a+b\,\text{ArcSin}[u])\,dx$ when u and $\int v\,dx$ are free of inverse functions

Derivation: Integration by parts

Rule: If u is free of inverse functions, let $w = \int v\,dx$, if w is free of inverse functions, then

$$\int v\,(a+b\,\text{ArcSin}[u])\,dx \longrightarrow w\,(a+b\,\text{ArcSin}[u]) - b\int \frac{w\,\partial_x u}{\sqrt{1-u^2}}\,dx$$

Program code:

```
Int[v_*(a_.+b_.*ArcSin[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcSin[u]),w,x] -
  b*Int[SimplifyIntegrand[w*D[u,x]/Sqrt[1-u^2],x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```

```
Int[v_*(a_.+b_.*ArcCos[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcCos[u]),w,x] +
  b*Int[SimplifyIntegrand[w*D[u,x]/Sqrt[1-u^2],x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```