Rules for integrands of the form $(d x)^m (a + b x^2 + c x^4)^p$

x.
$$\int (d x)^m (b x^2 + c x^4)^p dx$$

1: $\int (d x)^m (b x^2 + c x^4)^p dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If
$$p \in \mathbb{Z}$$
, then $(b x^2 + c x^4)^p = \frac{1}{d^{2p}} (d x)^{2p} (b + c x^2)^p$

Rule 1.2.2.2.0.1: If $p \in \mathbb{Z}$, then

$$\int \left(d\;x\right)^{\,m}\;\left(b\;x^2+c\;x^4\right)^{\,p}\;\mathrm{d}x\;\to\;\frac{1}{d^{2\;p}}\;\int \left(d\;x\right)^{\,m+2\;p}\;\left(b+c\;x^2\right)^{\,p}\;\mathrm{d}x$$

```
(* Int[(d_.*x_)^m_.*(b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
    1/d^(2*p)*Int[(d*x)^(m+2*p)*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,m},x] && IntegerQ[p] *)
```

2:
$$\int (dx)^m (bx^2 + cx^4)^p dx \text{ when } p \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_x \frac{(b x^2 + c x^4)^p}{(d x)^2 p (b + c x^2)^p} = 0$$

Rule 1.2.2.2.0.2: If $p \notin \mathbb{Z}$, then

$$\int \left(d\,x\right)^{m}\,\left(b\,x^{2}+c\,x^{4}\right)^{p}\,\mathrm{d}x\;\to\;\frac{\left(b\,x^{2}+c\,x^{4}\right)^{p}}{\left(d\,x\right)^{2\,p}\,\left(b+c\,x^{2}\right)^{p}}\int \left(d\,x\right)^{m+2\,p}\,\left(b+c\,x^{2}\right)^{p}\,\mathrm{d}x$$

```
(* Int[(d_.*x_)^m_.*(b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
   (b*x^2+c*x^4)^p/((d*x)^(2*p)*(b+c*x^2)^p)*Int[(d*x)^(m+2*p)*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,m,p},x] && Not[IntegerQ[p]] *)
```

1:
$$\int x (a + b x^2 + c x^4)^p dx$$

Derivation: Integration by substitution

Basis:
$$x F[x^2] = \frac{1}{2} Subst[F[x], x, x^2] \partial_x x^2$$

Rule 1.2.2.2.1:

$$\int x (a + b x^{2} + c x^{4})^{p} dx \rightarrow \frac{1}{2} Subst \left[\int (a + b x + c x^{2})^{p} dx, x, x^{2} \right]$$

Program code:

```
Int[x_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
    1/2*Subst[Int[(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x]
```

2:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx \text{ when } p \in \mathbb{Z}^+$$

Derivation: Algebraic expansion

Rule 1.2.2.2: If $p \in \mathbb{Z}^+$, then

$$\int \left(d\ x\right)^{m} \left(a+b\ x^{2}+c\ x^{4}\right)^{p} \, dx \ \longrightarrow \ \int ExpandIntegrand \left[\left(d\ x\right)^{m} \left(a+b\ x^{2}+c\ x^{4}\right)^{p},\ x\right] \, dx$$

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
   Int[ExpandIntegrand[(d*x)^m*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[p,0] && Not[IntegerQ[(m+1)/2]]
```

Derivation: Algebraic simplification

Basis: If
$$b^2 - 4$$
 a c == 0, then $a + b z + c z^2 = \frac{1}{c} (\frac{b}{2} + c z)^2$

Rule 1.2.2.2.3.1: If $b^2 - 4$ a $c = 0 \land p \in \mathbb{Z}$, then

$$\int \left(d\,x\right)^m\,\left(a+b\,x^2+c\,x^4\right)^p\,\mathrm{d}x\ \to\ \frac{1}{c^p}\int \left(d\,x\right)^m\,\left(\frac{b}{2}+c\,x^2\right)^{2\,p}\,\mathrm{d}x$$

```
(* Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
    1/c^p*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

Derivation: Square trinomial recurrence 2c with m + 4 p + 5 == 0

Rule 1.2.2.3.2.1: If
$$b^2 - 4$$
 a $c = 0 \land p \notin \mathbb{Z} \land m + 4p + 5 = 0 \land p < -1$, then

$$\int \left(d\;x\right)^{m}\;\left(a+b\;x^{2}+c\;x^{4}\right)^{p}\;\text{d}x\;\;\longrightarrow\;\;\frac{2\;\left(d\;x\right)^{m+1}\;\left(a+b\;x^{2}+c\;x^{4}\right)^{p+1}}{d\;\left(m+3\right)\;\left(2\;a+b\;x^{2}\right)}\;-\;\frac{\left(d\;x\right)^{m+1}\;\left(a+b\;x^{2}+c\;x^{4}\right)^{p+1}}{2\;a\;d\;\left(m+3\right)\;\left(p+1\right)}$$

2:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx$$
 when $b^2 - 4 a c = 0 \land p \notin \mathbb{Z} \land m + 4 p + 5 = 0 \land p \neq -\frac{1}{2}$

Derivation: Square trinomial recurrence 2c with m + 4 p + 5 == 0

Rule 1.2.2.3.2.1: If
$$b^2 - 4$$
 a $c = 0 \land p \notin \mathbb{Z} \land m + 4p + 5 = 0 \land p \neq -\frac{1}{2}$, then

$$\int \left(d\,x\right)^{m}\,\left(a+b\,x^{2}+c\,x^{4}\right)^{p}\,\mathrm{d}x\ \longrightarrow\ \frac{\left(d\,x\right)^{m+1}\,\left(a+b\,x^{2}+c\,x^{4}\right)^{p+1}}{4\,a\,d\,\left(p+1\right)\,\left(2\,p+1\right)}-\frac{\left(d\,x\right)^{m+1}\,\left(2\,a+b\,x^{2}\right)\,\left(a+b\,x^{2}+c\,x^{4}\right)^{p}}{4\,a\,d\,\left(2\,p+1\right)}$$

Program code:

?:
$$\int x^m (a + b x^2 + c x^4)^p dx$$
 when $b^2 - 4 a c = 0 \land p - \frac{1}{2} \in \mathbb{Z} \land \frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If
$$\frac{m-1}{2} \in \mathbb{Z}$$
, then $x^m \, F[x^2] = \frac{1}{2} \, \text{Subst}[x^{\frac{m-1}{2}} \, F[x], \, x, \, x^2] \, \partial_x x^2$

Rule 1.2.2.5.1: If
$$b^2-4$$
 a $c=0 \ \land \ p-\frac{1}{2}\in \mathbb{Z} \ \land \ \frac{m-1}{2}\in \mathbb{Z}$, then

$$\int \! x^m \, \left(a + b \, x^2 + c \, x^4 \right)^p \, \text{d} x \, \, \longrightarrow \, \, \frac{1}{2} \, \text{Subst} \Big[\int \! x^{\frac{m-1}{2}} \, \left(a + b \, x + c \, x^2 \right)^p \, \text{d} x \, , \, x \, , \, x^2 \Big]$$

```
Int[x_^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    1/2*Subst[Int[x^((m-1)/2)*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2] && IntegerQ[(m-1)/2] && (GtQ[m,0] || LtQ[0,4*p,-m-1])
```

Derivation: Piecewise constant extraction

Basis: If
$$b^2 - 4$$
 a $c = 0$, then $\partial_x \frac{\left(a + b \, x^2 + c \, x^4\right)^{p+1}}{\left(\frac{b}{2} + c \, x^2\right)^{2 \, (p+1)}} = 0$

Rule 1.2.2.3.2.2: If
$$b^2-4$$
 a $c=0 \ \land \ p-\frac{1}{2}\in \mathbb{Z} \ \land \ m\in \mathbb{Z}^+$, then

$$\int \left(d\;x\right)^{m}\;\left(a+b\;x^{2}+c\;x^{4}\right)^{p}\;\text{d}x\;\;\to\;\;\frac{c\;\left(a+b\;x^{2}+c\;x^{4}\right)^{p+1}}{\left(\frac{b}{2}+c\;x^{2}\right)^{2\;(p+1)}}\;\int \left(d\;x\right)^{m}\;\left(\frac{b}{2}+c\;x^{2}\right)^{2\;p}\;\text{d}x$$

```
(* Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    c*(a+b*x^2+c*x^4)^(p+1)/(b/2+c*x^2)^(2*(p+1))*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2] && IGeQ[m,2*p] *)
```

3:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx$$
 when $b^2 - 4 a c == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If
$$b^2 - 4$$
 a $c = 0$, then $\partial_x \frac{(a+b x^2+c x^4)^p}{(\frac{b}{2}+c x^2)^{2p}} = 0$

Note: If
$$b^2 - 4$$
 a c == 0, then $a + b z + c z^2 = \frac{1}{c} (\frac{b}{2} + c z)^2$

Rule 1.2.2.3.2.2: If $b^2 - 4$ a $c = 0 \land p \notin \mathbb{Z}$, then

$$\int \left(d\,x\right)^m\,\left(a+b\,x^2+c\,x^4\right)^p\,\text{d}x \ \to \ \frac{\left(a+b\,x^2+c\,x^4\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]}\,\left(\frac{b}{2}+c\,x^2\right)^{2\,\text{FracPart}[p]}}\,\int \left(d\,x\right)^m\,\left(\frac{b}{2}+c\,x^2\right)^{2\,p}\,\text{d}x$$

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    (a+b*x^2+c*x^4)^FracPart[p]/(c^IntPart[p]*(b/2+c*x^2)^(2*FracPart[p]))*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2]

Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/(1+2*c*x^2/b)^(2*FracPart[p])*Int[(d*x)^m*(1+2*c*x^2/b)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```

4:
$$\int x^m \left(a + b x^2 + c x^4\right)^p dx \text{ when } \frac{m-1}{2} \in \mathbb{Z}$$

Derivation: Integration by substitution

Basis: If
$$\frac{m-1}{2} \in \mathbb{Z}$$
, then $x^m F[x^2] = \frac{1}{2} \operatorname{Subst}[x^{\frac{m-1}{2}} F[x], x, x^2] \partial_x x^2$

Rule 1.2.2.2.5.1: If
$$\frac{m-1}{2} \in \mathbb{Z}$$
, then

$$\int x^{m} \left(a + b x^{2} + c x^{4} \right)^{p} dx \rightarrow \frac{1}{2} Subst \left[\int x^{\frac{m-1}{2}} \left(a + b x + c x^{2} \right)^{p} dx, x, x^{2} \right]$$

```
Int[x_^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
    1/2*Subst[Int[x^((m-1)/2)*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x] && IntegerQ[(m-1)/2]
```

5:
$$\int (dx)^m (a + bx^2 + cx^4)^p dx$$
 when $b^2 - 4ac \neq 0 \land m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If
$$k \in \mathbb{Z}^+$$
, then $(dx)^m F[x] = \frac{k}{d} \operatorname{Subst}[x^{k (m+1)-1} F[\frac{x^k}{d}], x, (dx)^{1/k}] \partial_x (dx)^{1/k}$

Rule 1.2.2.2.6.1.2: If $b^2 - 4$ a $c \neq 0 \land m \in \mathbb{F}$, let k = Denominator[m], then

```
Int[(d_.*x_)^m_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    With[{k=Denominator[m]},
    k/d*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(2*k)/d^2+c*x^(4*k)/d^4)^p,x],x,(d*x)^(1/k)]] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && FractionQ[m] && IntegerQ[p]
```

Derivation: Trinomial recurrence 1b with A = 0, B = 1 and m = m - n

Rule 1.2.2.2.6.1.3.1: If $b^2 - 4$ a $c \neq 0 \land p > 0 \land m > 1$, then

$$\begin{split} \int \left(d\ x\right)^m \ \left(a + b\ x^2 + c\ x^4\right)^p \, dx \ \longrightarrow \\ \frac{d\ \left(d\ x\right)^{m-1} \ \left(a + b\ x^2 + c\ x^4\right)^p \ \left(2\ b\ p + c\ (m+4\ p-1)\ x^2\right)}{c\ (m+4\ p+1)\ (m+4\ p-1)} \ - \\ \frac{2\ p\ d^2}{c\ (m+4\ p+1)\ \left(m+4\ p-1\right)} \ \int \left(d\ x\right)^{m-2} \ \left(a + b\ x^2 + c\ x^4\right)^{p-1} \ \left(a\ b\ (m-1) - \left(2\ a\ c\ (m+4\ p-1)\ - b^2\ (m+2\ p-1)\right)\ x^2\right) \, dx \end{split}$$

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    d*(d*x)^(m-1)*(a+b*x^2+c*x^4)^p*(2*b*p+c*(m+4*p-1)*x^2)/(c*(m+4*p+1)*(m+4*p-1)) -
    2*p*d^2/(c*(m+4*p+1)*(m+4*p-1))*
    Int[(d*x)^(m-2)*(a+b*x^2+c*x^4)^(p-1)*Simp[a*b*(m-1)-(2*a*c*(m+4*p-1)-b^2*(m+2*p-1))*x^2,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && GtQ[m,1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (d x)^m (a + b x^2 + c x^4)^p dx$ when $b^2 - 4 a c \neq 0 \land p > 0 \land m < -1$

Reference: G&R 2.160.2

Derivation: Trinomial recurrence 1a with A = 1 and B = 0

Rule 1.2.2.2.6.1.3.2: If $b^2 - 4$ a $c \neq 0 \land p > 0 \land m < -1$, then

3: $\int (d x)^m (a + b x^2 + c x^4)^p dx$ when $b^2 - 4 a c \neq 0 \land p > 0 \land m + 4 p + 1 \neq 0$

Derivation: Trinomial recurrence 1a with A = 0, B = 1 and m = m - n

Derivation: Trinomial recurrence 1b with A = 1 and B = 0

Rule 1.2.2.2.6.1.3.4: If $b^2 - 4$ a c $\neq 0 \land p > 0 \land m + 4p + 1 \neq 0$, then

Program code:

7.
$$\left(\left(d \ x \right)^m \left(a + b \ x^2 + c \ x^4 \right)^p dl x \text{ when } b^2 - 4 \ a \ c \neq 0 \ \land \ p < -1 \right)$$

1.
$$\left(\left(d \ x \right)^m \left(a + b \ x^2 + c \ x^4 \right)^p \ dx \ \text{ when } b^2 - 4 \ a \ c \neq 0 \ \land \ p < -1 \ \land \ m > 1 \right)$$

1:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx$$
 when $b^2 - 4 a c \neq 0 \land p < -1 \land 1 < m \le 3$

Derivation: Trinomial recurrence 2a with A = 1 and B = 0

Derivation: Trinomial recurrence 2b with A = 0, B = 1 and m = m - n

Rule 1.2.2.2.6.1.4.1.1: If $b^2 - 4$ a $c \neq 0 \land p < -1 \land 1 < m \leq 3$, then

$$\int (d x)^m (a + b x^2 + c x^4)^p dx \rightarrow$$

$$\frac{d \left(d \ x\right)^{m-1} \left(b+2 \ c \ x^2\right) \left(a+b \ x^2+c \ x^4\right)^{p+1}}{2 \ (p+1) \ \left(b^2-4 \ a \ c\right)} - \frac{d^2}{2 \ (p+1) \ \left(b^2-4 \ a \ c\right)} \int \left(d \ x\right)^{m-2} \left(b \ (m-1) + 2 \ c \ (m+4 \ p+5) \ x^2\right) \left(a+b \ x^2+c \ x^4\right)^{p+1} \ dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    d*(d*x)^(m-1)*(b+2*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*(p+1)*(b^2-4*a*c)) -
    d^2/(2*(p+1)*(b^2-4*a*c))*Int[(d*x)^(m-2)*(b*(m-1)+2*c*(m+4*p+5)*x^2)*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,1] && LeQ[m,3] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx$$
 when $b^2 - 4 a c \neq 0 \land p < -1 \land m > 3$

Derivation: Trinomial recurrence 2a with A = 0, B = 1 and m = m - n

Rule 1.2.2.2.6.1.4.1.2: If $b^2 - 4$ a c $\neq 0 \land p < -1 \land m > 3$, then

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    -d^3*(d*x)^(m-3)*(2*a+b*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*(p+1)*(b^2-4*a*c)) +
    d^4/(2*(p+1)*(b^2-4*a*c))*Int[(d*x)^(m-4)*(2*a*(m-3)+b*(m+4*p+3)*x^2)*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,3] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx$$
 when $b^2 - 4 a c \neq 0 \land p < -1$

Derivation: Trinomial recurrence 2b with A = 1 and B = 0

Rule 1.2.2.2.6.1.4.2: If $b^2 - 4$ a c $\neq 0 \land p < -1$, then

$$\int \left(d\ x\right)^m \left(a+b\ x^2+c\ x^4\right)^p \, \mathrm{d}x \ \longrightarrow \\ -\frac{\left(d\ x\right)^{m+1} \, \left(b^2-2\ a\ c+b\ c\ x^2\right) \, \left(a+b\ x^2+c\ x^4\right)^{p+1}}{2\ a\ d\ (p+1) \, \left(b^2-4\ a\ c\right)} + \\ \frac{1}{2\ a\ (p+1) \, \left(b^2-4\ a\ c\right)} \int \left(d\ x\right)^m \, \left(a+b\ x^2+c\ x^4\right)^{p+1} \, \left(b^2\ (m+2\ p+3)\ -2\ a\ c\ (m+4\ p+5)\ +b\ c\ (m+4\ p+7)\ x^2\right) \, \mathrm{d}x$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    -(d*x)^(m+1)*(b^2-2*a*c+b*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*a*d*(p+1)*(b^2-4*a*c)) +
    1/(2*a*(p+1)*(b^2-4*a*c))*
    Int[(d*x)^m*(a+b*x^2+c*x^4)^(p+1)*Simp[b^2*(m+2*p+3)-2*a*c*(m+4*p+5)+b*c*(m+4*p+7)*x^2,x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

8:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx$$
 when $b^2 - 4 a c \neq 0 \land m > 3 \land m + 4 p + 1 \neq 0$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with A = 0, B = 1 and m = m - n

Note: G&R 2.174.1 is a special case of G&R 2.160.3.

Rule 1.2.2.2.6.1.5: If $b^2 - 4$ a $c \neq 0 \land m > 3 \land m + 4 p + 1 \neq 0$, then

$$\int (d x)^m (a + b x^2 + c x^4)^p dx \rightarrow$$

$$\frac{d^{3} \left(d \ x\right)^{m-3} \left(a + b \ x^{2} + c \ x^{4}\right)^{p+1}}{c \ (m+4p+1)} - \frac{d^{4}}{c \ (m+4p+1)} \int \left(d \ x\right)^{m-4} \left(a \ (m-3) + b \ (m+2p-1) \ x^{2}\right) \left(a + b \ x^{2} + c \ x^{4}\right)^{p} dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    d^3*(d*x)^(m-3)*(a+b*x^2+c*x^4)^(p+1)/(c*(m+4*p+1)) -
    d^4/(c*(m+4*p+1))*
    Int[(d*x)^(m-4)*Simp[a*(m-3)+b*(m+2*p-1)*x^2,x]*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && GtQ[m,3] && NeQ[m+4*p+1,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

9: $\left((d x)^m (a + b x^2 + c x^4)^p dx \text{ when } b^2 - 4 a c \neq 0 \land m < -1 \right)$

Reference: G&R 2.160.1

Derivation: Trinomial recurrence 3b with A = 1 and B = 0

Note: G&R 2.161.6 is a special case of G&R 2.160.1.

Rule 1.2.2.2.6.1.6: If $b^2 - 4$ a $c \neq 0 \land m < -1$, then

```
Int[(d_.*x_)^m_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(a*d*(m+1)) -
  1/(a*d^2*(m+1))*Int[(d*x)^(m+2)*(b*(m+2*p+3)+c*(m+4*p+5)*x^2)*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && LtQ[m,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

10.
$$\int \frac{\left(d x\right)^{m}}{a + b x^{2} + c x^{4}} dx \text{ when } b^{2} - 4 a c \neq 0$$
1:
$$\int \frac{\left(d x\right)^{m}}{a + b x^{2} + c x^{4}} dx \text{ when } b^{2} - 4 a c \neq 0 \land m < -1$$

Reference: G&R 2.176, CRC 123

Derivation: Algebraic expansion

Basis:
$$\frac{(dz)^m}{a+bz+cz^2} = \frac{(dz)^m}{a} - \frac{1}{ad} \frac{(dz)^{m+1} (b+cz)}{a+bz+cz^2}$$

Rule 1.2.2.2.6.1.7.1: If $b^2 - 4$ a $c \neq 0 \land m < -1$, then

$$\int \frac{\left(d\;x\right)^{\,m}}{a+b\;x^{2}+c\;x^{4}}\;\mathrm{d}x\;\to\;\frac{\left(d\;x\right)^{\,m+1}}{a\;d\;\left(m+1\right)}-\frac{1}{a\;d^{2}}\int \frac{\left(d\;x\right)^{\,m+2}\;\left(b+c\;x^{2}\right)}{a+b\;x^{2}+c\;x^{4}}\;\mathrm{d}x$$

2.
$$\int \frac{\left(d\ x\right)^{m}}{a+b\ x^{2}+c\ x^{4}}\ dx \ \text{ when } b^{2}-4\ a\ c\neq 0\ \land\ m>3$$
1:
$$\int \frac{x^{m}}{a+b\ x^{2}+c\ x^{4}}\ dx \ \text{ when } b^{2}-4\ a\ c\neq 0\ \land\ m>5\ \land\ m\in\mathbb{Z}$$

Rule 1.2.2.2.6.1.7.2.1: If $\,b^2-4\,\,a\,\,c\,\neq\,0\,\,\wedge\,\,m\,>\,5\,\,\wedge\,\,m\,\in\,\mathbb{Z}$, then

$$\int \frac{x^{m}}{a+b x^{2}+c x^{4}} dx \rightarrow \int Polynomial Divide[x^{m}, a+b x^{2}+c x^{4}, x] dx$$

```
Int[x_^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
   Int[PolynomialDivide[x^m,(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && IGtQ[m,5]
```

2:
$$\int \frac{\left(d x\right)^{m}}{a+b x^{2}+c x^{4}} dx \text{ when } b^{2}-4 a c \neq 0 \land m > 3$$
 Not necessary?

Reference: G&R 2.174.1, CRC 119

Derivation: Algebraic expansion

Basis:
$$\frac{(d z)^m}{a+b z+c z^2} = \frac{d^2 (d z)^{m-2}}{c} - \frac{d^2}{c} \frac{(d z)^{m-2} (a+b z)}{a+b z+c z^2}$$

Rule 1.2.2.2.6.1.7.2.2: If $b^2 - 4$ a c $\neq 0 \land m > 3$, then

$$\int \frac{\left(d\ x\right)^{m}}{a+b\ x^{2}+c\ x^{4}}\ dx \ \to \ \frac{d^{3}\ \left(d\ x\right)^{m-3}}{c\ (m-3)} - \frac{d^{4}}{c} \int \frac{\left(d\ x\right)^{m-4}\ \left(a+b\ x^{2}\right)}{a+b\ x^{2}+c\ x^{4}}\ dx$$

```
 Int [ (d_{*}x_{-})^{m}/(a_{+}b_{*}x_{-}^{2}+c_{*}x_{-}^{4}), x_{-}Symbol ] := \\ d^{3}*(d*x)^{(m-3)}/(c*(m-3)) - d^{4}/c*Int[(d*x)^{(m-4)}*(a+b*x^{2})/(a+b*x^{2}+c*x^{4}), x] /; \\ FreeQ[\{a,b,c,d\},x] && NeQ[b^{2}-4*a*c,0] && GtQ[m,3]
```

3.
$$\int \frac{x^m}{a + b \, x^2 + c \, x^4} \, dx \text{ when } b^2 - 4 \, a \, c \neq 0 \ \land \ m \in \mathbb{Z}^+ \land \ 1 \leq m < 4 \ \land \ b^2 - 4 \, a \, c \, \neq 0$$

$$1: \int \frac{x^2}{a + b \, x^2 + c \, x^4} \, dx \text{ when } b^2 - 4 \, a \, c < 0 \ \land \ a \, c > 0$$

Basis: Let
$$q \to \sqrt{\frac{a}{c}}$$
, then $\frac{x^2}{a+b \ x^2+c \ x^4} = \frac{q+x^2}{2 \ (a+b \ x^2+c \ x^4)} - \frac{q-x^2}{2 \ (a+b \ x^2+c \ x^4)}$

Note: Resulting integrands are of the form $\frac{d+e \ x^2}{a+b \ x^2+c \ x^4}$ where c d² - a e² == 0 \land b² - 4 a c $\not>$ 0, for which there is rule.

Rule 1.2.2.2.6.1.7.3.1: If
$$\,b^2$$
 – 4 a c $<$ 0 $\,\wedge\,$ a c $>$ 0, let q $\rightarrow\,\sqrt{\frac{a}{c}}$, then

$$\int \frac{x^2}{a+b \ x^2+c \ x^4} \ \mathrm{d}x \ \to \ \frac{1}{2} \int \frac{q+x^2}{a+b \ x^2+c \ x^4} \ \mathrm{d}x - \frac{1}{2} \int \frac{q-x^2}{a+b \ x^2+c \ x^4} \ \mathrm{d}x$$

2:
$$\int \frac{x^m}{a + b \ x^2 + c \ x^4} \ dx \text{ when } b^2 - 4 \ a \ c \neq 0 \ \land \ m \in \mathbb{Z}^+ \land \ 3 \leq m < 4 \ \land \ b^2 - 4 \ a \ c \not \geqslant 0$$

Basis: If
$$q \to \sqrt{\frac{a}{c}}$$
 and $r \to \sqrt{2 \, q - \frac{b}{c}}$, then $\frac{z^3}{a + b \, z^2 + c \, z^4} = \frac{q + r \, z}{2 \, c \, r \, \left(q + r \, z + z^2\right)} - \frac{q - r \, z}{2 \, c \, r \, \left(q - r \, z + z^2\right)}$

Note: If $(a \mid b \mid c) \in \mathbb{R} \land b^2 - 4 \ a \ c < 0$, then $\frac{a}{c} > 0$ and $2\sqrt{\frac{a}{c}} - \frac{b}{c} > 0$.

Rule 1.2.2.2.6.1.7.3.2: If
$$b^2 - 4$$
 a $c \neq 0$ \wedge $m \in \mathbb{Z}^+ \wedge 3 \leq m < 4$ \wedge $b^2 - 4$ a $c \not > 0$, let $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2\,q - \frac{b}{c}}$, then
$$\int \frac{x^m}{a + b\,x^2 + c\,x^4} \, \mathrm{d}x \, \to \, \frac{1}{2\,c\,r} \int \frac{x^{m-3}\,(q + r\,x)}{q + r\,x + x^2} \, \mathrm{d}x - \frac{1}{2\,c\,r} \int \frac{x^{m-3}\,(q - r\,x)}{q - r\,x + x^2} \, \mathrm{d}x$$

3:
$$\int \frac{x^m}{a+b \ x^2+c \ x^4} \ dx \ \text{ when } b^2-4 \ a \ c \neq 0 \ \land \ m \in \mathbb{Z}^+ \land \ 1 \leq m < 3 \ \land \ b^2-4 \ a \ c \neq 0$$

$$\text{Basis: If } q \rightarrow \sqrt{\frac{\underline{a}}{c}} \text{ and } r \rightarrow \sqrt{2\,q - \frac{\underline{b}}{c}} \text{ , then } \frac{\underline{z}}{a + b\;z^2 + c\;z^4} \ = \ \frac{1}{2\,c\;r\;\left(q - r\;z + z^2\right)} \ - \ \frac{1}{2\,c\;r\;\left(q + r\;z + z^2\right)}$$

Note: If $(a \mid b \mid c) \in \mathbb{R} \land b^2 - 4 \ a \ c < 0$, then $\frac{a}{c} > 0$ and 2 $\sqrt{\frac{a}{c}} - \frac{b}{c} > 0$.

Rule 1.2.2.2.6.1.7.3.3: If
$$b^2 - 4$$
 a $c \neq 0$ \wedge $m \in \mathbb{Z}^+ \wedge 1 \leq m < 3$ \wedge $b^2 - 4$ a $c \not > 0$, let $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2\,q - \frac{b}{c}}$, then
$$\int \frac{x^m}{a + b\,x^2 + c\,x^4} \, \mathrm{d}x \, \to \, \frac{1}{2\,c\,r} \int \frac{x^{m-1}}{q - r\,x + x^2} \, \mathrm{d}x - \frac{1}{2\,c\,r} \int \frac{x^{m-1}}{q + r\,x + x^2} \, \mathrm{d}x$$

```
Int[x_^m_./(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
With[{q=Rt[a/c,2]},
With[{r=Rt[2*q-b/c,2]},
    1/(2*c*r)*Int[x^(m-1)/(q-r*x+x^2),x] - 1/(2*c*r)*Int[x^(m-1)/(q+r*x+x^2),x]]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && GeQ[m,1] && LtQ[m,3] && NegQ[b^2-4*a*c]
```

4:
$$\int \frac{(d x)^m}{a + b x^2 + c x^4} dx \text{ when } b^2 - 4 a c \neq 0 \land m \geq 2$$

Reference: G&R 2.161.1a & G&R 2.161.3

Derivation: Algebraic expansion

Basis: Let
$$q \to \sqrt{b^2 - 4 \ a \ c}$$
 , then $\frac{(d \ z)^m}{a + b \ z + c \ z^2} = \frac{d}{2} \ \left(\frac{b}{q} + 1 \right) \ \frac{(d \ z)^{m-1}}{\frac{b}{2} + \frac{q}{2} + c \ z} - \frac{d}{2} \ \left(\frac{b}{q} - 1 \right) \ \frac{(d \ z)^{m-1}}{\frac{b}{2} - \frac{q}{2} + c \ z}$

Rule 1.2.2.2.6.1.7.4: If b^2-4 a c $\neq 0 \ \land \ m \geq 2$, let $q \rightarrow \sqrt{b^2-4}$ a c , then

$$\int \frac{\left(d\;x\right)^{m}}{a+b\;x^{2}+c\;x^{4}}\;\mathrm{d}x\;\;\to\;\;\frac{d^{2}}{2}\;\left(\frac{b}{q}+1\right)\int \frac{\left(d\;x\right)^{m-2}}{\frac{b}{2}+\frac{q}{2}+c\;x^{2}}\;\mathrm{d}x\;-\;\frac{d^{2}}{2}\;\left(\frac{b}{q}-1\right)\int \frac{\left(d\;x\right)^{m-2}}{\frac{b}{2}-\frac{q}{2}+c\;x^{2}}\;\mathrm{d}x$$

```
Int[(d_.*x_)^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
    With[{q=Rt[b^2-4*a*c,2]},
    d^2/2*(b/q+1)*Int[(d*x)^(m-2)/(b/2+q/2+c*x^2),x] -
    d^2/2*(b/q-1)*Int[(d*x)^(m-2)/(b/2-q/2+c*x^2),x]] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GeQ[m,2]
```

5:
$$\int \frac{(d x)^m}{a + b x^2 + c x^4} dx \text{ when } b^2 - 4 a c \neq 0$$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let
$$q \to \sqrt{b^2 - 4} \ a \ c$$
 , then $\frac{1}{a+b \ z+c \ z^2} = \frac{c}{q} \ \frac{1}{\frac{b}{2} - \frac{q}{2} + c \ z} - \frac{c}{q} \ \frac{1}{\frac{b}{2} + \frac{q}{2} + c \ z}$

Rule 1.2.2.2.6.1.7.5: If $\,b^2\,-\,4\,\,a\,\,c\,\neq\,0,$ let $q\,\rightarrow\,\sqrt{\,b^2\,-\,4\,\,a\,\,c\,}$, then

$$\int \frac{(d x)^{m}}{a + b x^{2} + c x^{4}} dx \rightarrow \frac{c}{q} \int \frac{(d x)^{m}}{\frac{b}{2} - \frac{q}{2} + c x^{2}} dx - \frac{c}{q} \int \frac{(d x)^{m}}{\frac{b}{2} + \frac{q}{2} + c x^{2}} dx$$

Program code:

11.
$$\int \frac{x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, dx \text{ when } b^2 - 4 \, a \, c \neq 0$$
1.
$$\int \frac{x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, dx \text{ when } b^2 - 4 \, a \, c > 0$$
1.
$$\int \frac{x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, dx \text{ when } b^2 - 4 \, a \, c > 0 \, \land \, c < 0$$

Derivation: Algebraic expansion

Basis: If
$$b^2-4$$
 a $c>0$ \wedge $c<0$, let $q\to\sqrt{b^2-4}$ a c , then

$$\sqrt{a + b x^2 + c x^4} = \frac{1}{2\sqrt{-c}} \sqrt{b + q + 2 c x^2} \sqrt{-b + q - 2 c x^2}$$

Rule 1.2.2.2.6.1.8.1.1: If $\,b^2-4$ a c $>0\,\,\wedge\,\,c<0,$ let $q\to\sqrt{b^2-4}$ a c $\,$, then

$$\int \frac{x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, \mathrm{d}x \, \, \rightarrow \, 2 \, \sqrt{-\,c} \, \, \int \frac{x^2}{\sqrt{b + q + 2 \, c \, x^2}} \, \sqrt{-\,b + q - 2 \, c \, x^2} \, \, \mathrm{d}x$$

Program code:

2.
$$\int \frac{x^2}{\sqrt{a + b \ x^2 + c \ x^4}} \ dx \ \text{ when } b^2 - 4 \ a \ c > 0 \ \land \ c \not < 0$$
1:
$$\int \frac{x^2}{\sqrt{a + b \ x^2 + c \ x^4}} \ dx \ \text{ when } b^2 - 4 \ a \ c > 0 \ \land \ \frac{c}{a} > 0 \ \land \ \frac{b}{a} < 0$$

Derivation: Algebraic expansion

$$\text{Rule 1.2.2.2.6.1.8.1.2.1: If } b^2 - 4 \text{ a } c > 0 \text{ } \wedge \text{ } \frac{c}{a} > 0 \text{ } \wedge \text{ } \frac{b}{a} < 0 \text{, let } q \rightarrow \sqrt{\frac{c}{a}} \text{ , then } \\ \int \frac{x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, \mathrm{d}x \, \rightarrow \, \frac{1}{q} \int \frac{1}{\sqrt{a + b \, x^2 + c \, x^4}} \, \mathrm{d}x - \frac{1}{q} \int \frac{1 - q \, x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, \mathrm{d}x$$

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
    With[{q=Rt[c/a,2]},
    1/q*Int[1/Sqrt[a+b*x^2+c*x^4],x] - 1/q*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && GtQ[c/a,0] && LtQ[b/a,0]
```

2:
$$\int \frac{x^2}{\sqrt{a+b \ x^2+c \ x^4}} \ dx \ \text{ when } b^2-4 \ a \ c>0 \ \land \ a<0 \ \land \ c>0$$

Rule 1.2.2.2.6.1.8.1.2.2: If $\,b^2$ – 4 a c > 0 $\,\wedge\,$ a < 0 $\,\wedge\,$ c > 0, let q \rightarrow $\sqrt{b^2$ – 4 a c $}$, then

$$\int \frac{x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,\mathrm{d}x \ \to \ -\frac{b-q}{2\,c}\,\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,\mathrm{d}x + \frac{1}{2\,c}\,\int \frac{b-q+2\,c\,x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,\mathrm{d}x$$

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
    With[{q=Rt[b^2-4*a*c,2]},
    -(b-q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[a,0] && GtQ[c,0]
```

Reference: G&R 3.153.1+

Rule 1.2.2.2.6.1.8.1.2.3.1: If b^2-4 a c>0, let $q\to\sqrt{b^2-4}$ a c , if $\frac{b+q}{a}>0$, then

$$\int \frac{x^2}{\sqrt{a+b\,x^2+c\,x^4}} \, \mathrm{d}x \ \to \ \frac{x\,\left(b+q+2\,c\,x^2\right)}{2\,c\,\sqrt{a+b\,x^2+c\,x^4}} - \frac{\sqrt{\frac{b+q}{2\,a}}\,\left(2\,a+\left(b+q\right)\,x^2\right)\,\sqrt{\frac{2\,a+\left(b-q\right)\,x^2}{2\,a+\left(b+q\right)\,x^2}}}{2\,c\,\sqrt{a+b\,x^2+c\,x^4}} \, \text{EllipticE}\Big[\text{ArcTan}\Big[\sqrt{\frac{b+q}{2\,a}}\,\,x\Big]\,,\,\, \frac{2\,q}{b+q}\Big]$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
    With[{q=Rt[b^2-4*a*c,2]},
    x*(b+q+2*c*x^2)/(2*c*Sqrt[a+b*x^2+c*x^4]) -
    Rt[(b+q)/(2*a),2]*(2*a+(b+q)*x^2)*Sqrt[(2*a+(b-q)*x^2)/(2*a+(b+q)*x^2)]/(2*c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcTan[Rt[(b+q)/(2*a),2]*x],2*q/(b+q)] /;
    PosQ[(b+q)/a] && Not[PosQ[(b-q)/a] && SimplerSqrtQ[(b-q)/(2*a),(b+q)/(2*a)]]] /;
    FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2:
$$\int \frac{x^2}{\sqrt{a+b\,x^2+c\,x^4}} \, dx \text{ when } b^2-4\,a\,c>0 \ \land \ \frac{b-\sqrt{b^2-4\,a\,c}}{a}>0$$

Reference: G&R 3.153.1-

Rule 1.2.2.2.6.1.8.1.2.3.2: If b^2 – 4 a c > 0, let q $\rightarrow \sqrt{b^2$ – 4 a c , if $\frac{b-q}{a}$ > 0 then

$$\int \frac{x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,\mathrm{d}x \ \rightarrow \ \frac{x\,\left(b-q+2\,c\,x^2\right)}{2\,c\,\sqrt{a+b\,x^2+c\,x^4}} \ - \ \frac{\sqrt{\frac{b-q}{2\,a}}\,\left(2\,a+\left(b-q\right)\,x^2\right)\,\sqrt{\frac{2\,a+\left(b+q\right)\,x^2}{2\,a+\left(b-q\right)\,x^2}}}{2\,c\,\sqrt{a+b\,x^2+c\,x^4}} \ EllipticE\Big[ArcTan\Big[\sqrt{\frac{b-q}{2\,a}}\ x\Big]\,, \ -\frac{2\,q}{b-q}\Big]$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
    With[{q=Rt[b^2-4*a*c,2]},
    x*(b-q+2*c*x^2)/(2*c*Sqrt[a+b*x^2+c*x^4]) -
    Rt[(b-q)/(2*a),2]*(2*a+(b-q)*x^2)*Sqrt[(2*a+(b+q)*x^2)/(2*a+(b-q)*x^2)]/(2*c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcTan[Rt[(b-q)/(2*a),2]*x],-2*q/(b-q)] /;
    PosQ[(b-q)/a]] /;
    FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

4.
$$\int \frac{x^2}{\sqrt{a+b \ x^2+c \ x^4}} \ dx \ \text{ when } b^2-4 \ a \ c > 0 \ \land \ \frac{b \pm \sqrt{b^2-4 \ a \ c}}{a} \ \not > 0$$
1:
$$\int \frac{x^2}{\sqrt{a+b \ x^2+c \ x^4}} \ dx \ \text{ when } b^2-4 \ a \ c > 0 \ \land \ \frac{b + \sqrt{b^2-4 \ a \ c}}{a} \ \not > 0$$

Derivation: Algebraic expansion

$$\text{Rule 1.4.1.8.1.2.4.1: If } b^2 - 4 \text{ a c} > 0, \text{ let } q \to \sqrt{b^2 - 4 \text{ a c}}, \text{ if } \frac{b+q}{a} \not > 0 \text{ then } \\ \int \frac{x^2}{\sqrt{a+b \, x^2 + c \, x^4}} \, \mathrm{d} x \to -\frac{b+q}{2 \, c} \int \frac{1}{\sqrt{a+b \, x^2 + c \, x^4}} \, \mathrm{d} x + \frac{1}{2 \, c} \int \frac{b+q+2 \, c \, x^2}{\sqrt{a+b \, x^2 + c \, x^4}} \, \mathrm{d} x$$

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
    With[{q=Rt[b^2-4*a*c,2]},
    -(b+q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b+q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
    NegQ[(b+q)/a] && Not[NegQ[(b-q)/a] && SimplerSqrtQ[-(b-q)/(2*a),-(b+q)/(2*a)]]] /;
    FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2:
$$\int \frac{x^2}{\sqrt{a+b \ x^2+c \ x^4}} \ dx \ \text{ when } b^2-4 \ a \ c > 0 \ \land \ \frac{b-\sqrt{b^2-4 \ a \ c}}{a} \ \geqslant 0$$

$$\text{Rule 1.4.1.8.1.2.4.2: If } b^2 - 4 \text{ a c} > 0, \text{ let } q \rightarrow \sqrt{b^2 - 4 \text{ a c}}, \text{ if } \frac{b-q}{a} \not > 0 \text{ then } \\ \int \frac{x^2}{\sqrt{a+b \, x^2 + c \, x^4}} \, \mathrm{d} x \, \rightarrow \, -\frac{b-q}{2 \, c} \int \frac{1}{\sqrt{a+b \, x^2 + c \, x^4}} \, \mathrm{d} x + \frac{1}{2 \, c} \int \frac{b-q+2 \, c \, x^2}{\sqrt{a+b \, x^2 + c \, x^4}} \, \mathrm{d} x$$

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
    With[{q=Rt[b^2-4*a*c,2]},
    -(b-q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
    NegQ[(b-q)/a]] /;
FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2.
$$\int \frac{x^2}{\sqrt{a + b \ x^2 + c \ x^4}} \ dx \text{ when } b^2 - 4 \ a \ c \not\ge 0$$
1:
$$\int \frac{x^2}{\sqrt{a + b \ x^2 + c \ x^4}} \ dx \text{ when } b^2 - 4 \ a \ c \not= 0 \ \land \ \frac{c}{a} > 0$$

$$\text{Rule 1.2.2.2.6.1.8.2.1: If } b^2 - 4 \text{ a } c \neq 0 \text{ } \wedge \text{ } \frac{c}{a} > 0 \text{, let } q \rightarrow \sqrt{\frac{c}{a}} \text{ , then } \\ \int \frac{x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, \mathrm{d} x \text{ } \rightarrow \text{ } \frac{1}{q} \int \frac{1}{\sqrt{a + b \, x^2 + c \, x^4}} \, \mathrm{d} x - \frac{1}{q} \int \frac{1 - q \, x^2}{\sqrt{a + b \, x^2 + c \, x^4}} \, \mathrm{d} x$$

Program code:

2:
$$\int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx \text{ when } b^2 - 4 a c \neq 0 \land \frac{c}{a} > 0$$

Derivation: Piecewise constant extraction

Basis: If
$$q \to \sqrt{b^2 - 4 \ a \ c}$$
, then $\partial_x \frac{\sqrt{1 + \frac{2 \ c \ x^2}{b - q}} \ \sqrt{1 + \frac{2 \ c \ x^2}{b + q}}}{\sqrt{a + b \ x^2 + c \ x^4}} = 0$

Rule 1.2.2.2.6.1.8.2.2: If b^2-4 a c $\neq 0$ \wedge $\frac{c}{a} \not> 0$, let $q \to \sqrt{b^2-4}$ a c , then

$$\int \frac{x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,\mathrm{d}x \ \to \ \frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}}\,\int \frac{x^2}{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}\,\,\mathrm{d}x$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
With[{q=Rt[b^2-4*a*c,2]},
Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
    Int[x^2/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && NegQ[c/a]
```

12:
$$\int (d x)^m (a + b x^2 + c x^4)^p dx$$

Derivation: Piecewise constant extraction

Basis:
$$\partial_{X} \frac{\left(a+b \ x^{2}+c \ x^{4}\right)^{p}}{\left(1+\frac{2 c \ x^{2}}{b+\sqrt{b^{2}-4 \ a \ c}}\right)^{p} \left(1+\frac{2 c \ x^{2}}{b-\sqrt{b^{2}-4 \ a \ c}}\right)^{p}} == 0$$

Rule 1.2.2.2.10:

$$\int \left(d\;x\right)^m \left(a+b\;x^2+c\;x^4\right)^p \, \mathrm{d}x \; \to \; \frac{a^{\text{IntPart}[p]} \, \left(a+b\;x^2+c\;x^4\right)^{\text{FracPart}[p]}}{\left(1+\frac{2\,c\,x^2}{b+\sqrt{b^2-4\,a\,c}}\right)^{\text{FracPart}[p]}} \int \left(d\;x\right)^m \left(1+\frac{2\,c\,x^2}{b+\sqrt{b^2-4\,a\,c}}\right)^p \left(1+\frac{2\,c\,x^2}{b-\sqrt{b^2-4\,a\,c}}\right)^p \, \mathrm{d}x$$

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
    a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/
        ((1+2*c*x^2/(b+Rt[b^2-4*a*c,2]))^FracPart[p]*(1+2*c*x^2/(b-Rt[b^2-4*a*c,2]))^FracPart[p])*
        Int[(d*x)^m*(1+2*c*x^2/(b+Sqrt[b^2-4*a*c]))^p*(1+2*c*x^2/(b-Sqrt[b^2-4*a*c]))^p,x] /;
FreeQ[{a,b,c,d,m,p},x]
```

S:
$$\int u^m (a + b v^2 + c v^4)^p dx$$
 when $v == d + e x \wedge u == f v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If
$$u = f v$$
, then $\partial_x \frac{u^m}{v^m} = 0$

Rule 1.2.2.2.S: If $v == d + e \times \wedge u == f v$, then

$$\int\! u^m \, \left(a + b \, v^2 + c \, v^4\right)^p \, \mathrm{d}x \ \longrightarrow \ \frac{u^m}{e \, v^m} \, \mathsf{Subst} \Big[\int\! x^m \, \left(a + b \, x^2 + c \, x^4\right)^p \, \mathrm{d}x \, , \, \, x \, , \, \, v \Big]$$

```
Int[u_^m_.*(a_.+b_.*v_^2+c_.*v_^4)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a+b*x^2+c*x^(2*2))^p,x],x,v] /;
FreeQ[{a,b,c,m,p},x] && LinearPairQ[u,v,x]
```