Rules for integrands of the form $(g \tan[e + f x])^p (a + b \sin[e + f x])^m$

1. $\int (g \tan[e+f x])^p (a+b \sin[e+f x])^m \, dx$ when $a^2 - b^2 == 0$

1: $\int \frac{(g \tan[e+f x])^p}{a+b \sin[e+f x]} \, dx$ when $a^2 - b^2 == 0$

Derivation: Algebraic expansion

Basis: If $a^2 - b^2 == 0$, then $\frac{1}{a+b \sin[z]} == \frac{\sec[z]^2}{a} - \frac{\sec[z] \tan[z]}{b}$

Note: If $p == -1$, it is better to use the following substitution rule, since it results in a more continuous antiderivative.

Rule: If $a^2 - b^2 == 0$, then

$$\int \frac{(g \tan[e+f x])^p}{a+b \sin[e+f x]} \, dx \rightarrow \frac{1}{a} \int \sec[e+f x]^2 (g \tan[e+f x])^p \, dx - \frac{1}{b g} \int \sec[e+f x] (g \tan[e+f x])^{p+1} \, dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_./(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  1/a*Int[Sec[e+f*x]^2*(g*Tan[e+f*x])^p,x] - 1/(b*g)*Int[Sec[e+f*x]*(g*Tan[e+f*x])^(p+1),x] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && NeQ[p,-1]
```

2: $\int \tan[e+f x]^p (a+b \sin[e+f x])^m \, dx$ when $a^2 - b^2 == 0 \land \frac{p+1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{p+1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0$, then $\tan[e + f x]^p == \frac{b \cos[e+f x] (b \sin[e+f x])^p}{(a-b \sin[e+f x])^{\frac{p+1}{2}} (a+b \sin[e+f x])^{\frac{p+1}{2}}}$

Basis: $\cos[e + f x] F[b \sin[e + f x]] == \frac{1}{b f} \text{Subst}[F[x], x, b \sin[e + f x]] \, \partial_x (b \sin[e + f x])$

Rule: If $a^2 - b^2 == 0 \land \frac{p+1}{2} \in \mathbb{Z}$, then

$$\int \operatorname{Tan}[e+f x]^p\,(a+b\operatorname{Sin}[e+f x])^m\,dx \longrightarrow b\int \frac{\operatorname{Cos}[e+f x]\,(b\operatorname{Sin}[e+f x])^p\,(a+b\operatorname{Sin}[e+f x])^{m-\frac{p+1}{2}}}{(a-b\operatorname{Sin}[e+f x])^{\frac{p+1}{2}}}\,dx$$

$$\longrightarrow \frac{1}{f}\operatorname{Subst}\left[\int \frac{x^p\,(a+x)^{m-\frac{p+1}{2}}}{(a-x)^{\frac{p+1}{2}}}\,dx,\ x,\ b\operatorname{Sin}[e+f x]\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  1/f*Subst[Int[x^p*(a+x)^(m-(p+1)/2)/(a-x)^((p+1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[(p+1)/2]
```

3. $\int (g\operatorname{Tan}[e+f x])^p\,(a+b\operatorname{Sin}[e+f x])^m\,dx$ when $a^2-b^2=0\wedge m\in\mathbb{Z}$

1: $\int \operatorname{Tan}[e+f x]^p\,(a+b\operatorname{Sin}[e+f x])^m\,dx$ when $a^2-b^2=0\wedge m\in\mathbb{Z}\wedge p=2m$

Derivation: Algebraic simplification

Basis: If $a^2-b^2=0\wedge m\in\mathbb{Z}\wedge p=2m$, then $\operatorname{Tan}[e+f x]^p\,(a+b\operatorname{Sin}[e+f x])^m=\frac{a^p\operatorname{Sin}[e+f x]^p}{(a-b\operatorname{Sin}[e+f x])^m}$

Rule: If $a^2-b^2=0\wedge m\in\mathbb{Z}\wedge p=2m$, then

$$\int \operatorname{Tan}[e+f x]^p\,(a+b\operatorname{Sin}[e+f x])^m\,dx \longrightarrow a^p\int \frac{\operatorname{Sin}[e+f x]^p}{(a-b\operatorname{Sin}[e+f x])^m}\,dx$$

Program code:

```
Int[tan[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  a^p*Int[Sin[e+f*x]^p/(a-b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && IntegersQ[m,p] && EqQ[p,2*m]
```

2: $\int \mathrm{Tan}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,\left(m\mid\frac{p}{2}\right)\in\mathbb{Z}\,\land\,\left(p<0\lor m-\frac{p}{2}>0\right)$

Derivation: Algebraic expansion

Basis: If $a^2-b^2==0\,\land\,\frac{p}{2}\in\mathbb{Z}$, then $\mathrm{Tan}[e+f\,x]^p==\frac{a^p\,\mathrm{Sin}[e+f\,x]^p}{(a+b\,\mathrm{Sin}[e+f\,x])^{p/2}\,(a-b\,\mathrm{Sin}[e+f\,x])^{p/2}}$

Rule: If $a^2-b^2==0\,\land\,\left(m\mid\frac{p}{2}\right)\in\mathbb{Z}\,\land\,\left(p<0\lor m-\frac{p}{2}>0\right)$, then

$$\int \mathrm{Tan}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\;\longrightarrow\;a^p\int \mathrm{ExpandIntegrand}\left[\frac{\mathrm{Sin}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-\frac{p}{2}}}{(a-b\,\mathrm{Sin}[e+f\,x])^{p/2}},\,x\right]dx$$

Program code:

```
Int[tan[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  a^p*Int[ExpandIntegrand[Sin[e+f*x]^p*(a+b*Sin[e+f*x])^(m-p/2)/(a-b*Sin[e+f*x])^(p/2),x],x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && IntegersQ[m,p/2] && (LtQ[p,0] || GtQ[m-p/2,0])
```

3: $\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $a^2-b^2==0\,\land\,m\in\mathbb{Z}^+$, then

$$\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\;\longrightarrow\;\int (g\,\mathrm{Tan}[e+f\,x])^p\,\mathrm{ExpandIntegrand}\left[(a+b\,\mathrm{Sin}[e+f\,x])^m,\,x\right]dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  Int[ExpandIntegrand[(g*Tan[e+f*x])^p,(a+b*Sin[e+f*x])^m,x],x] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && IGtQ[m,0]
```

4: $\int (g \operatorname{Tan}[e + f x])^p (a + b \operatorname{Sin}[e + f x])^m \, dx$ when $a^2 - b^2 == 0 \land m \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $a^2 - b^2 == 0 \land m \in \mathbb{Z}$, then $(a + b \operatorname{Sin}[e + f x])^m == a^{2m} \operatorname{Sec}[e + f x]^{-m} (a \operatorname{Sec}[e + f x] - b \operatorname{Tan}[e + f x])^{-m}$

Rule: If $a^2 - b^2 == 0 \land m \in \mathbb{Z}^-$, then

$$\int (g \operatorname{Tan}[e + f x])^p (a + b \operatorname{Sin}[e + f x])^m \, dx \longrightarrow a^{2m} \int (g \operatorname{Tan}[e + f x])^p \operatorname{Sec}[e + f x]^{-m} \operatorname{ExpandIntegrand}\left[(a \operatorname{Sec}[e + f x] - b \operatorname{Tan}[e + f x])^{-m}, x\right] dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  a^(2*m)*Int[ExpandIntegrand[(g*Tan[e+f*x])^p*Sec[e+f*x]^(-m),(a*Sec[e+f*x]-b*Tan[e+f*x])^(-m),x],x] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && ILtQ[m,0]
```

4. $\int (g \operatorname{Tan}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \notin \mathbb{Z}$

1. $\int (g \operatorname{Tan}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \notin \mathbb{Z} \land \frac{p}{2} \in \mathbb{Z}$

1. $\int \operatorname{Tan}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \notin \mathbb{Z}$

1: $\int \operatorname{Tan}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \notin \mathbb{Z} \land m<0$

Derivation: ???

Rule: If $a^2-b^2=0 \land m \notin \mathbb{Z} \land m<0$, then

$$\int \operatorname{Tan}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$
$$\frac{b (a+b \operatorname{Sin}[e+f x])^m}{a f (2 m-1) \operatorname{Cos}[e+f x]}-\frac{1}{a^2 (2 m-1)} \int \frac{(a+b \operatorname{Sin}[e+f x])^{m+1} (a m-b (2 m-1) \operatorname{Sin}[e+f x])}{\operatorname{Cos}[e+f x]^2} \, dx$$

Program code:

```
Int[tan[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(a+b*Sin[e+f*x])^m/(a*f*(2*m-1)*Cos[e+f*x]) -
  1/(a^2*(2*m-1))*Int[(a+b*Sin[e+f*x])^(m+1)*(a*m-b*(2*m-1)*Sin[e+f*x])/Cos[e+f*x]^2,x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && LtQ[m,0]
```

2: $\int \operatorname{Tan}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \notin \mathbb{Z} \land m \nless 0$

Derivation: Nondegenerate sine recurrence 1b with $A \to a^2$, $B \to 2 a b$, $C \to b^2$, $m \to 0$

Rule: If $a^2-b^2=0 \land m \notin \mathbb{Z} \land m \nless 0$, then

$$\int \operatorname{Tan}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$

$$-\frac{(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,m\,\mathrm{Cos}[e+f\,x]}+\frac{1}{b\,m}\int\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(b\,(m+1)+a\,\mathrm{Sin}[e+f\,x])}{\mathrm{Cos}[e+f\,x]^2}\,dx$$

Program code:

```
Int[tan[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -(a+b*Sin[e+f*x])^(m+1)/(b*f*m*Cos[e+f*x]) +
  1/(b*m)*Int[(a+b*Sin[e+f*x])^m*(b*(m+1)+a*Sin[e+f*x])/Cos[e+f*x]^2,x] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && Not[LtQ[m,0]]
```

2: $\int \mathrm{Tan}[e+f\,x]^4\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\wedge m-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\mathrm{Tan}[z]^4==1-\frac{1-2\,\mathrm{Sin}[z]^2}{\mathrm{Cos}[z]^4}$

Rule: If $a^2-b^2==0\wedge m-\frac{1}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Tan}[e+f\,x]^4\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\,\rightarrow\,\int(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx-\int\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,\left(1-2\,\mathrm{Sin}[e+f\,x]^2\right)}{\mathrm{Cos}[e+f\,x]^4}\,dx$$

Program code:

```
Int[tan[e_.+f_.*x_]^4*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Int[(a+b*Sin[e+f*x])^m,x] - Int[(a+b*Sin[e+f*x])^m*(1-2*Sin[e+f*x]^2)/Cos[e+f*x]^4,x] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[m-1/2]
```

3. $\int\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^2}\,dx$ when $a^2-b^2==0\wedge m-\frac{1}{2}\in\mathbb{Z}$

1: $\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^2}\,dx$ when $a^2-b^2==0\ \land\ m-\frac{1}{2}\in\mathbb{Z}\ \land\ m<-1$

Rule: If $a^2-b^2==0\ \land\ m-\frac{1}{2}\in\mathbb{Z}\ \land\ m<-1$, then

$$\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^2}\,dx \longrightarrow -\frac{(a+b\,\text{Sin}[e+f\,x])^{m+1}}{a\,f\,\text{Tan}[e+f\,x]}+\frac{1}{b^2}\int \frac{(a+b\,\text{Sin}[e+f\,x])^{m+1}\,(b\,m-a\,(m+1)\,\text{Sin}[e+f\,x])}{\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^2,x_Symbol] :=
  -(a+b*Sin[e+f*x])^(m+1)/(a*f*Tan[e+f*x]) +
  1/b^2*Int[(a+b*Sin[e+f*x])^(m+1)*(b*m-a*(m+1)*Sin[e+f*x])/Sin[e+f*x],x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && IntegerQ[m-1/2] && LtQ[m,-1]
```

2: $\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^2}\,dx$ when $a^2-b^2==0\ \land\ m-\frac{1}{2}\in\mathbb{Z}\ \land\ m\not<-1$

Rule: If $a^2-b^2==0\ \land\ m-\frac{1}{2}\in\mathbb{Z}\ \land\ m\not<-1$, then

$$\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^2}\,dx \longrightarrow -\frac{(a+b\,\text{Sin}[e+f\,x])^{m}}{f\,\text{Tan}[e+f\,x]}+\frac{1}{a}\int \frac{(a+b\,\text{Sin}[e+f\,x])^{m}\,(b\,m-a\,(m+1)\,\text{Sin}[e+f\,x])}{\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_./tan[e_.+f_.*x_]^2,x_Symbol] :=
  -(a+b*Sin[e+f*x])^m/(f*Tan[e+f*x]) +
  1/a*Int[(a+b*Sin[e+f*x])^m*(b*m-a*(m+1)*Sin[e+f*x])/Sin[e+f*x],x] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[m-1/2] && Not[LtQ[m,-1]]
```

4. $\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^4}\,dx$ when $a^2-b^2==0\land m-\frac{1}{2}\in\mathbb{Z}$

1: $\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^4}\,dx$ when $a^2-b^2==0\land m-\frac{1}{2}\in\mathbb{Z}\land m<-1$

Derivation: Algebraic expansion

Basis: If $a^2-b^2==0$, then $\frac{1}{\mathrm{Tan}[z]^4}==-\frac{2\,(a+b\,\mathrm{Sin}[z])^2}{a\,b\,\mathrm{Sin}[z]^3}+\frac{(a+b\,\mathrm{Sin}[z])^2\,(1+\mathrm{Sin}[z]^2)}{a^2\,\mathrm{Sin}[z]^4}$

Rule: If $a^2-b^2==0\land m-\frac{1}{2}\in\mathbb{Z}\land m<-1$, then

$$\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^4}\,dx\to-\frac{2}{a\,b}\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^{m+2}}{\mathrm{Sin}[e+f\,x]^3}\,dx+\frac{1}{a^2}\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^{m+2}\,(1+\mathrm{Sin}[e+f\,x]^2)}{\mathrm{Sin}[e+f\,x]^4}\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^4,x_Symbol] :=
  -2/(a*b)*Int[(a+b*Sin[e+f*x])^(m+2)/Sin[e+f*x]^3,x] +
  1/a^2*Int[(a+b*Sin[e+f*x])^(m+2)*(1+Sin[e+f*x]^2)/Sin[e+f*x]^4,x] /;
FreeQ[{a,b,e,f},x] && EqQ[a^2-b^2,0] && IntegerQ[m-1/2] && LtQ[m,-1]
```

2: $\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^4}\,dx$ when $a^2-b^2==0\ \land\ m-\frac{1}{2}\in\mathbb{Z}\ \land\ m\not<-1$

Derivation: Algebraic expansion

Basis: $\frac{1}{\mathrm{Tan}[z]^4}==1+\frac{1-2\,\mathrm{Sin}[z]^2}{\mathrm{Sin}[z]^4}$

Rule: If $a^2-b^2==0\ \land\ m-\frac{1}{2}\in\mathbb{Z}\ \land\ m\not<-1$, then

$$\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^4}\,dx\ \longrightarrow\ \int (a+b\,\mathrm{Sin}[e+f\,x])^m\,dx+\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(1-2\,\mathrm{Sin}[e+f\,x]^2)}{\mathrm{Sin}[e+f\,x]^4}\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^4,x_Symbol] :=
  Int[(a+b*Sin[e+f*x])^m,x] + Int[(a+b*Sin[e+f*x])^m*(1-2*Sin[e+f*x]^2)/Sin[e+f*x]^4,x] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && IntegerQ[m-1/2] && Not[LtQ[m,-1]]
```

5: $\int \mathrm{Tan}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\ \land\ m\notin\mathbb{Z}\ \land\ \frac{p}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0\ \land\ \frac{p}{2}\in\mathbb{Z}$, then $\mathrm{Tan}[e+f\,x]^p==\frac{(b\,\mathrm{Sin}[e+f\,x])^p}{(a-b\,\mathrm{Sin}[e+f\,x])^{p/2}\,(a+b\,\mathrm{Sin}[e+f\,x])^{p/2}}$

Basis: If $a^2-b^2==0$, then $\partial_x\frac{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}{\mathrm{Cos}[e+f\,x]}==0$

Basis: $\mathrm{Cos}[e+f\,x]\,F[b\,\mathrm{Sin}[e+f\,x]]==\frac{1}{b\,f}\,\mathrm{Subst}[F[x],x,b\,\mathrm{Sin}[e+f\,x]]\,\partial_x(b\,\mathrm{Sin}[e+f\,x])$

Rule: If $a^2-b^2==0\ \land\ m\notin\mathbb{Z}\ \land\ \frac{p}{2}\in\mathbb{Z}$, then

$$\int \text{Tan}[e+f x]^p\,(a+b\,\text{Sin}[e+f x])^m\,dx \to \int \frac{(b\,\text{Sin}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^{m-p/2}}{(a-b\,\text{Sin}[e+f x])^{p/2}}\,dx$$

$$\to \frac{\sqrt{a+b\,\text{Sin}[e+f x]}\,\sqrt{a-b\,\text{Sin}[e+f x]}}{\text{Cos}[e+f x]} \int \frac{\text{Cos}[e+f x]\,(b\,\text{Sin}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^{m-\frac{p+1}{2}}}{(a-b\,\text{Sin}[e+f x])^{\frac{p+1}{2}}}\,dx$$

$$\to \frac{\sqrt{a+b\,\text{Sin}[e+f x]}\,\sqrt{a-b\,\text{Sin}[e+f x]}}{b\,f\,\text{Cos}[e+f x]}\,\text{Subst}\Big[\int \frac{x^p\,(a+x)^{m-\frac{p+1}{2}}}{(a-x)^{\frac{p+1}{2}}}\,dx,\ x,\ b\,\text{Sin}[e+f x]\Big]$$

Program code:

```
Int[tan[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Sqrt[a+b*Sin[e+f*x]]*Sqrt[a-b*Sin[e+f*x]]/(b*f*Cos[e+f*x])*
    Subst[Int[x^p*(a+x)^(m-(p+1)/2)/(a-x)^((p+1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && IntegerQ[p/2]
```

2: $\int (g\,\text{Tan}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^m\,dx$ when $a^2-b^2=0\,\land\,m\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0$, then

$$\partial_x\left(\left((g\,\text{Tan}[e+f x])^{p+1}\,(a-b\,\text{Sin}[e+f x])^{\frac{p+1}{2}}\,(a+b\,\text{Sin}[e+f x])^{\frac{p+1}{2}}\right)\Big/(b\,\text{Sin}[e+f x])^{p+1}\right)=0$$

Basis: $\text{Cos}[e+f x]\,F[b\,\text{Sin}[e+f x]] = \frac{1}{b\,f}\,\text{Subst}[F[x],x,b\,\text{Sin}[e+f x]]\,\partial_x(b\,\text{Sin}[e+f x])$

Rule: If $a^2-b^2=0\,\land\,m\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$, then

$$\int (g\,\text{Tan}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^m\,dx \to$$

$$\to \frac{b\,(g\,\text{Tan}[e+f x])^{p+1}\,(a-b\,\text{Sin}[e+f x])^{\frac{p+1}{2}}\,(a+b\,\text{Sin}[e+f x])^{\frac{p+1}{2}}}{g\,(b\,\text{Sin}[e+f x])^{p+1}} \int \frac{\text{Cos}[e+f x]\,(b\,\text{Sin}[e+f x])^p\,(a+b\,\text{Sin}[e+f x])^{m-\frac{p+1}{2}}}{(a-b\,\text{Sin}[e+f x])^{\frac{p+1}{2}}}\,dx$$

$$\to \frac{(g\,\mathrm{Tan}[e+f\,x])^{p+1}\,(a-b\,\mathrm{Sin}[e+f\,x])^{\frac{p+1}{2}}\,(a+b\,\mathrm{Sin}[e+f\,x])^{\frac{p+1}{2}}}{f\,g\,(b\,\mathrm{Sin}[e+f\,x])^{p+1}}\,\mathrm{Subst}\left[\int\frac{x^p\,(a+x)^{m-\frac{p+1}{2}}}{(a-x)^{\frac{p+1}{2}}}\,dx,\,x,\,b\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (g*Tan[e+f*x])^(p+1)*(a-b*Sin[e+f*x])^((p+1)/2)*(a+b*Sin[e+f*x])^((p+1)/2)/(f*g*(b*Sin[e+f*x])^(p+1))*
    Subst[Int[x^p*(a+x)^(m-(p+1)/2)/(a-x)^((p+1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]] && Not[IntegerQ[p]]
```

2. $\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0$

1: $\int \mathrm{Tan}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\ \land\ \frac{p+1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{p+1}{2}\in\mathbb{Z}$, then $\mathrm{Tan}[e+f\,x]^p == \frac{b\,\mathrm{Cos}[e+f\,x]\,(b\,\mathrm{Sin}[e+f\,x])^p}{\left(b^2-b^2\,\mathrm{Sin}[e+f\,x]^2\right)^{\frac{p+1}{2}}}$

Basis: $\mathrm{Cos}[e+f\,x]\,F[b\,\mathrm{Sin}[e+f\,x]] == \frac{1}{b\,f}\,\mathrm{Subst}[F[x],x,b\,\mathrm{Sin}[e+f\,x]]\,\partial_x\,(b\,\mathrm{Sin}[e+f\,x])$

Rule: If $a^2-b^2\neq 0\ \land\ \frac{p+1}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Tan}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\ \longrightarrow\ b\int \frac{\mathrm{Cos}[e+f\,x]\,(b\,\mathrm{Sin}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m}{\left(b^2-b^2\,\mathrm{Sin}[e+f\,x]^2\right)^{\frac{p+1}{2}}}\,dx$$

$$\longrightarrow\ \frac{1}{f}\,\mathrm{Subst}\left[\int \frac{x^p\,(a+x)^m}{\left(b^2-x^2\right)^{\frac{p+1}{2}}}\,dx,\ x,\ b\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  1/f*Subst[Int[(x^p*(a+x)^m)/(b^2-x^2)^((p+1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0] && IntegerQ[(p+1)/2]
```

2: $\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\,\land\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $a^2-b^2\neq 0\,\land\,m\in\mathbb{Z}^+$, then

$$\int (g\,\mathrm{Tan}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\;\longrightarrow\;\int (g\,\mathrm{Tan}[e+f\,x])^p\,\mathrm{ExpandIntegrand}\left[(a+b\,\mathrm{Sin}[e+f\,x])^m,\,x\right]dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  Int[ExpandIntegrand[(g*Tan[e+f*x])^p,(a+b*Sin[e+f*x])^m,x],x] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

3. $\int (g \operatorname{Tan}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land \frac{p}{2} \in \mathbb{Z}$

1: $\int \frac{(a+b \operatorname{Sin}[e+f x])^m}{\operatorname{Tan}[e+f x]^2} \, dx$ when $a^2-b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\operatorname{Tan}[z]^2} = \frac{1-\operatorname{Sin}[z]^2}{\operatorname{Sin}[z]^2}$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{(a+b \operatorname{Sin}[e+f x])^m}{\operatorname{Tan}[e+f x]^2} \, dx \to \int \frac{(a+b \operatorname{Sin}[e+f x])^m \left(1-\operatorname{Sin}[e+f x]^2\right)}{\operatorname{Sin}[e+f x]^2} \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^2,x_Symbol] :=
  Int[(a+b*Sin[e+f*x])^m*(1-Sin[e+f*x]^2)/Sin[e+f*x]^2,x] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0]
```

2. $\int \frac{(a+b \operatorname{Sin}[e+f x])^m}{\operatorname{Tan}[e+f x]^4} \, dx$ when $a^2-b^2 \neq 0$

1: $\int \frac{(a+b \operatorname{Sin}[e+f x])^m}{\operatorname{Tan}[e+f x]^4} \, dx$ when $a^2-b^2 \neq 0 \land m < -1$

Derivation: Algebraic expansion

Basis: $\frac{1}{\operatorname{Tan}[z]^4} = 1 + \frac{1-2 \operatorname{Sin}[z]^2}{\operatorname{Sin}[z]^4}$

Rule: If $a^2-b^2 \neq 0 \land m < -1$, then

$$\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^4}\,dx \longrightarrow \int (a+b\,\text{Sin}[e+f\,x])^m\,dx + \int \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(1-2\,\text{Sin}[e+f\,x]^2)}{\text{Sin}[e+f\,x]^4}\,dx \longrightarrow$$

$$-\frac{\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{3\,a\,f\,\text{Sin}[e+f\,x]^3} - \frac{(3\,a^2+b^2\,(m-2))\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{3\,a^2\,b\,f\,(m+1)\,\text{Sin}[e+f\,x]^2} -$$

$$\frac{1}{3\,a^2\,b\,(m+1)}\int \frac{1}{\text{Sin}[e+f\,x]^3}\,(a+b\,\text{Sin}[e+f\,x])^{m+1}\,\left(6\,a^2-b^2\,(m-1)\,(m-2)+a\,b\,(m+1)\,\text{Sin}[e+f\,x]-(3\,a^2-b^2\,m\,(m-2))\,\text{Sin}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^4,x_Symbol] :=
  -Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(3*a*f*Sin[e+f*x]^3) -
  (3*a^2+b^2*(m-2))*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(3*a^2*b*f*(m+1)*Sin[e+f*x]^2) -
  1/(3*a^2*b*(m+1))*Int[(a+b*Sin[e+f*x])^(m+1)/Sin[e+f*x]^3*
    Simp[6*a^2-b^2*(m-1)*(m-2)+a*b*(m+1)*Sin[e+f*x]-(3*a^2-b^2*m*(m-2))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && IntegerQ[2*m]
```

x: $\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^4}\,dx$ when $a^2-b^2\neq 0\land m\nless -1$

Derivation: Algebraic expansion

Basis: $\frac{1}{\mathrm{Tan}[z]^4}=1+\frac{1-2\,\mathrm{Sin}[z]^2}{\mathrm{Sin}[z]^4}$

Rule: If $a^2-b^2\neq 0\land m\nless -1$, then

$$\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{\mathrm{Tan}[e+f\,x]^4}\,dx \longrightarrow \int (a+b\,\mathrm{Sin}[e+f\,x])^m\,dx+\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(1-2\,\mathrm{Sin}[e+f\,x]^2)}{\mathrm{Sin}[e+f\,x]^4}\,dx \longrightarrow$$

$$-\frac{\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{3\,a\,f\,\mathrm{Sin}[e+f\,x]^3}-\frac{\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,m\,\mathrm{Sin}[e+f\,x]^2}-$$

$$\frac{1}{3\,a\,b\,m}\int \frac{1}{\mathrm{Sin}[e+f\,x]^3}(a+b\,\mathrm{Sin}[e+f\,x])^m\,\left(6\,a^2-b^2\,m\,(m-2)+a\,b\,(m+3)\,\mathrm{Sin}[e+f\,x]-\left(3\,a^2-b^2\,m\,(m-1)\right)\,\mathrm{Sin}[e+f\,x]^2\right)\,dx$$

Program code:

```
(* Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^4,x_Symbol] :=
  -Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(3*a*f*Sin[e+f*x]^3) -
  Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*m*Sin[e+f*x]^2) -
  1/(3*a*b*m)*Int[(a+b*Sin[e+f*x])^m/Sin[e+f*x]^3*
    Simp[6*a^2-b^2*m*(m-2)+a*b*(m+3)*Sin[e+f*x]-(3*a^2-b^2*m*(m-1))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]] && IntegerQ[2*m] *)
```

2: $\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^4}\,dx$ when $a^2-b^2\neq 0\wedge m\not<-1$

Basis: $\frac{1}{\text{Tan}[z]^4}==\frac{1}{\text{Sin}[z]^4}-\frac{2-\text{Sin}[z]^2}{\text{Sin}[z]^2}$

Rule: If $a^2-b^2\neq 0\wedge m\not<-1$, then

$$\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^4}\,dx\to\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Sin}[e+f\,x]^4}\,dx-\int \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(2-\text{Sin}[e+f\,x]^2)}{\text{Sin}[e+f\,x]^2}\,dx\to$$

$$-\frac{\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{3\,a\,f\,\text{Sin}[e+f\,x]^3}-\frac{b\,(m-2)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{6\,a^2\,f\,\text{Sin}[e+f\,x]^2}-$$

$$\frac{1}{6\,a^2}\int \frac{1}{\text{Sin}[e+f\,x]^2}(a+b\,\text{Sin}[e+f\,x])^m\left(8\,a^2-b^2\,(m-1)\,(m-2)+a\,b\,m\,\text{Sin}[e+f\,x]-\left(6\,a^2-b^2\,m\,(m-2)\right)\text{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^4,x_Symbol] :=
  -Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(3*a*f*Sin[e+f*x]^3) -
  b*(m-2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(6*a^2*f*Sin[e+f*x]^2) -
  1/(6*a^2)*Int[(a+b*Sin[e+f*x])^m/Sin[e+f*x]^2*
    Simp[8*a^2-b^2*(m-1)*(m-2)+a*b*m*Sin[e+f*x]-(6*a^2-b^2*m*(m-2))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]] && IntegerQ[2*m]
```

3: $\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^6}\,dx$ when $a^2-b^2\neq 0\wedge m\neq 1$

Basis: $\frac{1}{\text{Tan}[z]^6}==\frac{1-3\,\text{Sin}[z]^2}{\text{Sin}[z]^6}+\frac{3-\text{Sin}[z]^2}{\text{Sin}[z]^2}$

Rule: If $a^2-b^2\neq 0\wedge m\neq 1$, then

$$\int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Tan}[e+f\,x]^6}\,dx \to \int \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(1-3\,\text{Sin}[e+f\,x]^2)}{\text{Sin}[e+f\,x]^6}\,dx + \int \frac{(a+b\,\text{Sin}[e+f\,x])^m\,(3-\text{Sin}[e+f\,x]^2)}{\text{Sin}[e+f\,x]^2}\,dx \to$$

$$-\frac{\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{5\,a\,f\,\text{Sin}[e+f\,x]^5} - \frac{b\,(m-4)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{20\,a^2\,f\,\text{Sin}[e+f\,x]^4} +$$

$$\frac{a\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{b^2\,f\,m\,(m-1)\,\text{Sin}[e+f\,x]^3} + \frac{\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{b\,f\,m\,\text{Sin}[e+f\,x]^2} + \frac{1}{20\,a^2\,b^2\,m\,(m-1)} \int \frac{(a+b\,\text{Sin}[e+f\,x])^m}{\text{Sin}[e+f\,x]^4} \cdot$$

$$\left(60\,a^4 - 44\,a^2\,b^2\,(m-1)\,m + b^4\,m\,(m-1)\,(m-3)\,(m-4) +\right.$$
$$\left. a\,b\,m\,\left(20\,a^2 - b^2\,m\,(m-1)\right)\,\text{Sin}[e+f\,x] - \left(40\,a^4 + b^4\,m\,(m-1)\,(m-2)\,(m-4) - 20\,a^2\,b^2\,(m-1)\,(2\,m+1)\right)\,\text{Sin}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_/tan[e_.+f_.*x_]^6,x_Symbol] :=
  -Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(5*a*f*Sin[e+f*x]^5) -
  b*(m-4)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(20*a^2*f*Sin[e+f*x]^4) +
  a*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b^2*f*m*(m-1)*Sin[e+f*x]^3) +
  Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*m*Sin[e+f*x]^2) +
  1/(20*a^2*b^2*m*(m-1))*Int[(a+b*Sin[e+f*x])^m/Sin[e+f*x]^4*
     Simp[60*a^4-44*a^2*b^2*(m-1)*m+b^4*m*(m-1)*(m-3)*(m-4)+a*b*m*(20*a^2-b^2*m*(m-1))*Sin[e+f*x]-
       (40*a^4+b^4*m*(m-1)*(m-2)*(m-4)-20*a^2*b^2*(m-1)*(2*m+1))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,m},x] && NeQ[a^2-b^2,0] && NeQ[m,1] && IntegerQ[2*m]
```

4. $\int \frac{(g\,\text{Tan}[e+f\,x])^p}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2 \neq 0 \land 2\,p \in \mathbb{Z}$

1: $\int \frac{(g\,\text{Tan}[e+f\,x])^p}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2 \neq 0 \land 2\,p \in \mathbb{Z} \land p > 1$

Derivation: Algebraic expansion

Basis: $\frac{\text{Tan}[z]^2}{a+b\,\text{Sin}[z]} == \frac{a\,\text{Tan}[z]^2}{(a^2-b^2)\,\text{Sin}[z]^2} - \frac{b\,\text{Tan}[z]}{(a^2-b^2)\,\text{Cos}[z]} - \frac{a^2}{(a^2-b^2)\,(a+b\,\text{Sin}[z])}$

Rule: If $a^2-b^2 \neq 0 \land 2\,p \in \mathbb{Z} \land p > 1$, then

$$\int \frac{(g\,\text{Tan}[e+f\,x])^p}{a+b\,\text{Sin}[e+f\,x]}\,dx \to \frac{a}{a^2-b^2}\int \frac{(g\,\text{Tan}[e+f\,x])^p}{\text{Sin}[e+f\,x]^2}\,dx - \frac{b\,g}{a^2-b^2}\int \frac{(g\,\text{Tan}[e+f\,x])^{p-1}}{\text{Cos}[e+f\,x]}\,dx - \frac{a^2\,g^2}{a^2-b^2}\int \frac{(g\,\text{Tan}[e+f\,x])^{p-2}}{a+b\,\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  a/(a^2-b^2)*Int[(g*Tan[e+f*x])^p/Sin[e+f*x]^2,x] -
  b*g/(a^2-b^2)*Int[(g*Tan[e+f*x])^(p-1)/Cos[e+f*x],x] -
  a^2*g^2/(a^2-b^2)*Int[(g*Tan[e+f*x])^(p-2)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*p] && GtQ[p,1]
```

2: $\int \frac{(g\,\text{Tan}[e+f\,x])^p}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\wedge\,2\,p\in\mathbb{Z}\,\wedge\,p<-1$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\,\text{Sin}[z]} == \frac{1}{a\,\text{Cos}[z]^2} - \frac{b\,\text{Tan}[z]}{a^2\,\text{Cos}[z]} - \frac{(a^2-b^2)\,\text{Tan}[z]^2}{a^2\,(a+b\,\text{Sin}[z])}$

Rule: If $a^2-b^2\neq 0\,\wedge\,2\,p\in\mathbb{Z}\,\wedge\,p<-1$, then

$$\int \frac{(g\,\text{Tan}[e+f\,x])^p}{a+b\,\text{Sin}[e+f\,x]}\,dx \to \frac{1}{a}\int \frac{(g\,\text{Tan}[e+f\,x])^p}{\text{Cos}[e+f\,x]^2}\,dx - \frac{b}{a^2\,g}\int \frac{(g\,\text{Tan}[e+f\,x])^{p+1}}{\text{Cos}[e+f\,x]}\,dx - \frac{a^2-b^2}{a^2\,g^2}\int \frac{(g\,\text{Tan}[e+f\,x])^{p+2}}{a+b\,\text{Sin}[e+f\,x]}\,dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  1/a*Int[(g*Tan[e+f*x])^p/Cos[e+f*x]^2,x] -
  b/(a^2*g)*Int[(g*Tan[e+f*x])^(p+1)/Cos[e+f*x],x] -
  (a^2-b^2)/(a^2*g^2)*Int[(g*Tan[e+f*x])^(p+2)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*p] && LtQ[p,-1]
```

3: $\int \frac{\sqrt{g\,\mathrm{Tan}[e+f\,x]}}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{\mathrm{Cos}[e+f\,x]}\,\sqrt{g\,\mathrm{Tan}[e+f\,x]}}{\sqrt{\mathrm{Sin}[e+f\,x]}} == 0$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{\sqrt{g\,\mathrm{Tan}[e+f\,x]}}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx \longrightarrow \frac{\sqrt{\mathrm{Cos}[e+f\,x]}\,\sqrt{g\,\mathrm{Tan}[e+f\,x]}}{\sqrt{\mathrm{Sin}[e+f\,x]}}\int \frac{\sqrt{\mathrm{Sin}[e+f\,x]}}{\sqrt{\mathrm{Cos}[e+f\,x]}\,(a+b\,\mathrm{Sin}[e+f\,x])}\,dx$$

Program code:

```
Int[Sqrt[g_.*tan[e_.+f_.*x_]]/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  Sqrt[Cos[e+f*x]]*Sqrt[g*Tan[e+f*x]]/Sqrt[Sin[e+f*x]]*Int[Sqrt[Sin[e+f*x]]/(Sqrt[Cos[e+f*x]]*(a+b*Sin[e+f*x])),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

4: $\int \frac{1}{\sqrt{g\,\text{Tan}[e+f\,x]}\,(a+b\,\text{Sin}[e+f\,x])}\,dx$ when $a^2-b^2\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{\text{Sin}[e+f\,x]}}{\sqrt{\text{Cos}[e+f\,x]}\,\sqrt{g\,\text{Tan}[e+f\,x]}} == 0$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{1}{\sqrt{g\,\text{Tan}[e+f\,x]}\,(a+b\,\text{Sin}[e+f\,x])}\,dx \to \frac{\sqrt{\text{Sin}[e+f\,x]}}{\sqrt{\text{Cos}[e+f\,x]}\,\sqrt{g\,\text{Tan}[e+f\,x]}}\int \frac{\sqrt{\text{Cos}[e+f\,x]}}{\sqrt{\text{Sin}[e+f\,x]}\,(a+b\,\text{Sin}[e+f\,x])}\,dx$$

Program code:

```
Int[1/(Sqrt[g_*tan[e_.+f_.*x_]]*(a_+b_.*sin[e_.+f_.*x_])),x_Symbol] :=
  Sqrt[Sin[e+f*x]]/(Sqrt[Cos[e+f*x]]*Sqrt[g*Tan[e+f*x]])*Int[Sqrt[Cos[e+f*x]]/(Sqrt[Sin[e+f*x]]*(a+b*Sin[e+f*x])),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

5: $\int \text{Tan}[e+f\,x]^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0\,\land\,\left(m\mid\frac{p}{2}\right)\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $\frac{p}{2}\in\mathbb{Z}$, then $\text{Tan}[e+f\,x]^p == \frac{\text{Sin}[e+f\,x]^p}{\left(1-\text{Sin}[e+f\,x]^2\right)^{p/2}}$

Rule: If $a^2-b^2\neq 0\,\land\,\left(m\mid\frac{p}{2}\right)\in\mathbb{Z}$, then

$$\int \text{Tan}[e+f\,x]^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{\text{Sin}[e+f\,x]^p\,(a+b\,\text{Sin}[e+f\,x])^m}{\left(1-\text{Sin}[e+f\,x]^2\right)^{p/2}},\,x\right]dx$$

Program code:

```
Int[tan[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Int[ExpandIntegrand[Sin[e+f*x]^p*(a+b*Sin[e+f*x])^m/(1-Sin[e+f*x]^2)^(p/2),x],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a^2-b^2,0] && IntegersQ[m,p/2]
```

X: $\int (g\,\text{Tan}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$

Rule:

$$\int (g\,\text{Tan}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx \longrightarrow \int (g\,\text{Tan}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$$

Program code:

```
Int[(g_.*tan[e_.+f_.*x_])^p_.*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  Unintegrable[(g*Tan[e+f*x])^p*(a+b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,g,m,p},x]
```

Rules for integrands of the form $(g\,\mathrm{Cot}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m$

1: $\int (g\,\mathrm{Cot}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((g\,\mathrm{Cot}[e+f\,x])^p\,(g\,\mathrm{Tan}[e+f\,x])^p\right) == 0$

Rule: If $p\notin\mathbb{Z}$, then

$$\int (g\,\mathrm{Cot}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx \longrightarrow g^{2\,\mathrm{IntPart}[p]}\,(g\,\mathrm{Cot}[e+f\,x])^{\mathrm{FracPart}[p]}\,(g\,\mathrm{Tan}[e+f\,x])^{\mathrm{FracPart}[p]}\int\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m}{(g\,\mathrm{Tan}[e+f\,x])^p}\,dx$$

Program code:

```
Int[(g_.*cot[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.,x_Symbol] :=
  g^(2*IntPart[p])*(g*Cot[e+f*x])^FracPart[p]*(g*Tan[e+f*x])^FracPart[p]*Int[(a+b*Sin[e+f*x])^m/(g*Tan[e+f*x])^p,x] /;
FreeQ[{a,b,e,f,g,m,p},x] && Not[IntegerQ[p]]
```