

## Rules for integrands of the form $\text{Sin}[a + b x + c x^2]^n$

1.  $\int \text{Sin}[a + b x + c x^2] \, dx$

1:  $\int \text{Sin}[a + b x + c x^2] \, dx$  when  $b^2 - 4 a c == 0$

Derivation: Algebraic simplification

Basis: If  $b^2 - 4 a c == 0$ , then  $a + b x + c x^2 == \frac{(b+2 c x)^2}{4 c}$

Rule: If  $b^2 - 4 a c == 0$ , then

$$\int \text{Sin}[a + b x + c x^2] \, dx \rightarrow \int \text{Sin}\left[\frac{(b+2 c x)^2}{4 c}\right] \, dx$$

Program code:

```
Int[Sin[a_+b_*x_+c_*x_^2],x_Symbol] :=
  Int[Sin[(b+2*c*x)^2/(4*c)],x] /;
  FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0]
```

```
Int[Cos[a_+b_*x_+c_*x_^2],x_Symbol] :=
  Int[Cos[(b+2*c*x)^2/(4*c)],x] /;
  FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0]
```

2:  $\int \text{Sin}[a + b x + c x^2] \, dx$  when  $b^2 - 4 a c \neq 0$

Derivation: Algebraic expansion

Basis:  $a + b x + c x^2 == \frac{(b+2 c x)^2}{4 c} - \frac{b^2-4 a c}{4 c}$

Basis:  $\text{Sin}[z - w] == \text{Cos}[w] \text{Sin}[z] - \text{Sin}[w] \text{Cos}[z]$

Rule: If  $b^2 - 4 a c \neq 0$ , then

$$\int \sin[a + b x + c x^2] dx \rightarrow \cos\left[\frac{b^2 - 4 a c}{4 c}\right] \int \sin\left[\frac{(b + 2 c x)^2}{4 c}\right] dx - \sin\left[\frac{b^2 - 4 a c}{4 c}\right] \int \cos\left[\frac{(b + 2 c x)^2}{4 c}\right] dx$$

Program code:

```
Int[Sin[a_.+b_.*x_.+c_.*x_^2],x_Symbol] :=
  Cos[(b^2-4*a*c)/(4*c)]*Int[Sin[(b+2*c*x)^2/(4*c)],x] -
  Sin[(b^2-4*a*c)/(4*c)]*Int[Cos[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

```
Int[Cos[a_.+b_.*x_.+c_.*x_^2],x_Symbol] :=
  Cos[(b^2-4*a*c)/(4*c)]*Int[Cos[(b+2*c*x)^2/(4*c)],x] +
  Sin[(b^2-4*a*c)/(4*c)]*Int[Sin[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

**2:**  $\int \sin[a + b x + c x^2]^n dx$  when  $n \in \mathbb{Z} \wedge n > 1$

Derivation: Algebraic expansion

Rule: If  $n \in \mathbb{Z} \wedge n > 1$ , then

$$\int \sin[a + b x + c x^2]^n dx \rightarrow \text{TrigReduce}[\sin[a + b x + c x^2]^n] dx$$

Program code:

```
Int[Sin[a_.+b_.*x_.+c_.*x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[Sin[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[n,1]
```

```
Int[Cos[a_.+b_.*x_.+c_.*x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[Cos[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[n,1]
```

**X:**  $\int \sin[a + b x + c x^2]^n dx$

Rule:

$$\int \sin[a + b x + c x^2]^n dx \rightarrow \int \sin[a + b x + c x^2]^n dx$$

Program code:

```
Int[Sin[a_.+b_.*x_.+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[Sin[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,n},x]
```

```
Int[Cos[a_.+b_.*x_.+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[Cos[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,n},x]
```

**N:**  $\int \sin[v]^n dx$  when  $n \in \mathbb{Z}^+ \wedge v == a + b x + c x^2$

Derivation: Algebraic normalization

Rule: If  $n \in \mathbb{Z}^+ \wedge v == a + b x + c x^2$ , then

$$\int \sin[v]^n dx \rightarrow \int \sin[a + b x + c x^2]^n dx$$

Program code:

```
Int[Sin[v_]^n_.,x_Symbol] :=
  Int[Sin[ExpandToSum[v,x]]^n,x] /;
IGtQ[n,0] && QuadraticQ[v,x] && Not[QuadraticMatchQ[v,x]]
```

```
Int[Cos[v_]^n_, x_Symbol] :=
  Int[Cos[ExpandToSum[v, x]]^n, x] /;
  IGtQ[n, 0] && QuadraticQ[v, x] && Not[QuadraticMatchQ[v, x]]
```

### Rules for integrands of the form $(d + e x)^m \sin[a + b x + c x^2]^n$

$$1. \int (d + e x)^m \sin[a + b x + c x^2] dx$$

$$1. \int (d + e x)^m \sin[a + b x + c x^2] dx \text{ when } 2 c d - b e == 0$$

$$\mathbf{1:} \int (d + e x) \sin[a + b x + c x^2] dx \text{ when } 2 c d - b e == 0$$

Derivation: Inverted integration by parts with  $m \rightarrow 1$

Rule: If  $2 c d - b e == 0$ , then

$$\int (d + e x) \sin[a + b x + c x^2] dx \rightarrow -\frac{e \cos[a + b x + c x^2]}{2 c}$$

Program code:

```
Int[(d_+e_.**x_)*Sin[a_+b_.**x_+c_.**x_^2], x_Symbol] :=
  -e**Cos[a+b**x+c**x^2]/(2*c) /;
  FreeQ[{a,b,c,d,e}, x] && EqQ[2*c*d-b*e, 0]
```

```
Int[(d_+e_.**x_)*Cos[a_+b_.**x_+c_.**x_^2], x_Symbol] :=
  e**Sin[a+b**x+c**x^2]/(2*c) /;
  FreeQ[{a,b,c,d,e}, x] && EqQ[2*c*d-b*e, 0]
```

$$\mathbf{2:} \int (d+e x)^m \sin[a+b x+c x^2] dx \text{ when } 2 c d-b e == 0 \wedge m > 1$$

Derivation: Inverted integration by parts

Rule: If  $2 c d-b e == 0 \wedge m > 1$ , then

$$\int (d+e x)^m \sin[a+b x+c x^2] dx \rightarrow -\frac{e (d+e x)^{m-1} \cos[a+b x+c x^2]}{2 c} + \frac{e^2 (m-1)}{2 c} \int (d+e x)^{m-2} \cos[a+b x+c x^2] dx$$

Program code:

```
Int[(d_+e_*x_)^m_*Sin[a_+b_*x_+c_*x_^2],x_Symbol] :=
  -e*(d+e*x)^(m-1)*Cos[a+b*x+c*x^2]/(2*c) +
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && GtQ[m,1]
```

```
Int[(d_+e_*x_)^m_*Cos[a_+b_*x_+c_*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sin[a+b*x+c*x^2]/(2*c) -
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && GtQ[m,1]
```

$$\mathbf{3:} \int (d+e x)^m \sin[a+b x+c x^2] dx \text{ when } 2 c d-b e == 0 \wedge m < -1$$

Derivation: Integration by parts

$$\text{Basis: } (d+e x)^m == \partial_x \frac{(d+e x)^{m+1}}{e (m+1)}$$

$$\text{Basis: If } 2 c d-b e == 0, \text{ then } \partial_x \sin[a+b x+c x^2] == \frac{2 c}{e} (d+e x) \cos[a+b x+c x^2]$$

Rule: If  $2 c d-b e == 0 \wedge m < -1$ , then

$$\int (d+e x)^m \sin[a+b x+c x^2] dx \rightarrow \frac{(d+e x)^{m+1} \sin[a+b x+c x^2]}{e(m+1)} - \frac{2c}{e^2(m+1)} \int (d+e x)^{m+2} \cos[a+b x+c x^2] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sin[a+b*x+c*x^2]/(e*(m+1)) -
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && LtQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cos[a+b*x+c*x^2]/(e*(m+1)) +
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && LtQ[m,-1]
```

2.  $\int (d+e x)^m \sin[a+b x+c x^2] dx$  when  $2 c d - b e \neq 0$

1:  $\int (d+e x) \sin[a+b x+c x^2] dx$  when  $2 c d - b e \neq 0$

Rule: If  $2 c d - b e \neq 0$ , then

$$\int (d+e x) \sin[a+b x+c x^2] dx \rightarrow -\frac{e \cos[a+b x+c x^2]}{2 c} + \frac{2 c d - b e}{2 c} \int \sin[a+b x+c x^2] dx$$

Program code:

```
Int[(d_.+e_.*x_)*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*Cos[a+b*x+c*x^2]/(2*c) +
  (2*c*d-b*e)/(2*c)*Int[Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0]
```

```
Int[(d_.+e_.*x_)*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Sin[a+b*x+c*x^2]/(2*c) +
  (2*c*d-b*e)/(2*c)*Int[Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0]
```

**2:**  $\int (d+e x)^m \sin[a+b x+c x^2] dx$  when  $b e - 2 c d \neq 0 \wedge m > 1$

Rule: If  $b e - 2 c d \neq 0 \wedge m > 1$ , then

$$\int (d+e x)^m \sin[a+b x+c x^2] dx \rightarrow$$

$$-\frac{e (d+e x)^{m-1} \cos[a+b x+c x^2]}{2 c} - \frac{b e - 2 c d}{2 c} \int (d+e x)^{m-1} \sin[a+b x+c x^2] dx + \frac{e^2 (m-1)}{2 c} \int (d+e x)^{m-2} \cos[a+b x+c x^2] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*(d+e*x)^(m-1)*Cos[a+b*x+c*x^2]/(2*c) -
  (b*e-2*c*d)/(2*c)*Int[(d+e*x)^(m-1)*Sin[a+b*x+c*x^2],x] +
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && GtQ[m,1]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sin[a+b*x+c*x^2]/(2*c) -
  (b*e-2*c*d)/(2*c)*Int[(d+e*x)^(m-1)*Cos[a+b*x+c*x^2],x] -
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && GtQ[m,1]
```

**3:**  $\int (d+e x)^m \sin[a+b x+c x^2] dx$  when  $b e - 2 c d \neq 0 \wedge m < -1$

Rule: If  $b e - 2 c d \neq 0 \wedge m < -1$ , then

$$\int (d+e x)^m \sin[a+b x+c x^2] dx \rightarrow \frac{(d+e x)^{m+1} \sin[a+b x+c x^2]}{e(m+1)} - \frac{b e-2 c d}{e^2(m+1)} \int (d+e x)^{m+1} \cos[a+b x+c x^2] dx - \frac{2 c}{e^2(m+1)} \int (d+e x)^{m+2} \cos[a+b x+c x^2] dx$$

Program code:

```
Int[(d_+e_.x_)^m_*Sin[a_+b_.x_+c_.x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sin[a+b*x+c*x^2]/(e*(m+1)) -
  (b*e-2*c*d)/(e^2*(m+1))*Int[(d+e*x)^(m+1)*Cos[a+b*x+c*x^2],x] -
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && LtQ[m,-1]
```

```
Int[(d_+e_.x_)^m_*Cos[a_+b_.x_+c_.x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cos[a+b*x+c*x^2]/(e*(m+1)) +
  (b*e-2*c*d)/(e^2*(m+1))*Int[(d+e*x)^(m+1)*Sin[a+b*x+c*x^2],x] +
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && LtQ[m,-1]
```

**2:**  $\int (d+e x)^m \sin[a+b x+c x^2]^n dx$  when  $n-1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If  $n-1 \in \mathbb{Z}^+$ , then

$$\int (d+e x)^m \sin[a+b x+c x^2]^n dx \rightarrow \int (d+e x)^m \text{TrigReduce}[\sin[a+b x+c x^2]^n] dx$$

Program code:

```
Int[(d_+e_.x_)^m_*Sin[a_+b_.x_+c_.x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[(d+e*x)^m,Sin[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,1]
```



```

Int[(d_+e_.x_)^m_.*Cos[a_+b_.x_+c_.x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[(d+e*x)^m,Cos[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,1]

```

**X:**  $\int (d+e x)^m \operatorname{Sin}[a+b x+c x^2]^n dx$

Rule:

$$\int (d+e x)^m \operatorname{Sin}[a+b x+c x^2]^n dx \rightarrow \int (d+e x)^m \operatorname{Sin}[a+b x+c x^2]^n dx$$

Program code:

```

Int[(d_+e_.x_)^m_.*Sin[a_+b_.x_+c_.x_^2]^n_,x_Symbol] :=
  Unintegrable[(d+e*x)^m*Sin[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]

```

```

Int[(d_+e_.x_)^m_.*Cos[a_+b_.x_+c_.x_^2]^n_,x_Symbol] :=
  Unintegrable[(d+e*x)^m*Cos[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]

```

**N:**  $\int u^m \sin[v]^n dx$  when  $n \in \mathbb{Z}^+ \wedge u = d + e x \wedge v = a + b x + c x^2$

Derivation: Algebraic normalization

Rule: If  $n \in \mathbb{Z}^+ \wedge u = d + e x \wedge v = a + b x + c x^2$ , then

$$\int u^m \sin[v]^n dx \rightarrow \int (d + e x)^m \sin[a + b x + c x^2]^n dx$$

Program code:

```
Int[u_^m_.*Sin[v_]^n_,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*Sin[ExpandToSum[v,x]^n,x] /;
  FreeQ[m,x] && IGtQ[n,0] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]]
```

```
Int[u_^m_.*Cos[v_]^n_,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*Cos[ExpandToSum[v,x]^n,x] /;
  FreeQ[m,x] && IGtQ[n,0] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]]
```