

Rules for integrands of the form $(a + b x + c x^2)^p$

1. $\int (a + b x + c x^2)^p dx$ when $b^2 - 4 a c == 0$

1: $\int (a + b x + c x^2)^p dx$ when $b^2 - 4 a c == 0 \wedge p < -1$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4 a c == 0$, then $\partial_x \frac{(a + b x + c x^2)^{p+1}}{(b + 2 c x)^{2 (p+1)}} == 0$

Rule 1.2.1.1.1.1: If $b^2 - 4 a c == 0 \wedge p < -1$, then

$$\int (a + b x + c x^2)^p dx \rightarrow \frac{4 c (a + b x + c x^2)^{p+1}}{(b + 2 c x)^{2 (p+1)}} \int (b + 2 c x)^{2 p} dx \rightarrow \frac{2 (a + b x + c x^2)^{p+1}}{(2 p + 1) (b + 2 c x)}$$

Program code:

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  2*(a+b*x+c*x^2)^(p+1)/((2*p+1)*(b+2*c*x)) /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && LtQ[p,-1]
```

2. $\int (a + b x + c x^2)^p dx$ when $b^2 - 4 a c == 0 \wedge p \neq -1$

1: $\int \frac{1}{\sqrt{a + b x + c x^2}} dx$ when $b^2 - 4 a c == 0$

Reference: G&R 2.261.3 which is correct only for $\frac{b}{2} + c x > 0$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4 a c == 0$, then $\partial_x \frac{\frac{b}{2} + c x}{\sqrt{a + b x + c x^2}} == 0$

Rule 1.2.1.1.1.1: If $b^2 - 4 a c == 0$, then

$$\int \frac{1}{\sqrt{a+bx+cx^2}} dx \rightarrow \frac{\frac{b}{2} + cx}{\sqrt{a+bx+cx^2}} \int \frac{1}{\frac{b}{2} + cx} dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*x_+c_.*x_^2],x_Symbol] :=
  (b/2+c*x)/Sqrt[a+b*x+c*x^2]*Int[1/(b/2+c*x),x] /;
FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0]
```

2: $\int (a+bx+cx^2)^p dx$ when $b^2 - 4ac == 0 \wedge p \neq -\frac{1}{2}$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4ac == 0$, then $\partial_x \frac{(a+bx+cx^2)^p}{(b+2cx)^{2p}} == 0$

Rule 1.2.1.1.1.2: If $b^2 - 4ac == 0 \wedge p \neq -\frac{1}{2}$, then

$$\int (a+bx+cx^2)^p dx \rightarrow \frac{(a+bx+cx^2)^p}{(b+2cx)^{2p}} \int (b+2cx)^{2p} dx \rightarrow \frac{(b+2cx)(a+bx+cx^2)^p}{2c(2p+1)}$$

Program code:

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^p/(2*c*(2*p+1)) /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && NeQ[p,-1/2]
```

$$2. \int (a + bx + cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge 4p \in \mathbb{Z} \wedge p > 0$$

$$1. \int (a + bx + cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge p > 0 \wedge p \in \mathbb{Z}$$

$$\mathbf{1:} \int (a + bx + cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge p \in \mathbb{Z}^+ \wedge \text{PerfectSquare}[b^2 - 4ac]$$

Derivation: Algebraic expansion

Basis: Let $q = \sqrt{b^2 - 4ac}$, then $a + bz + cz^2 = \frac{1}{c} \left(\frac{b}{2} - \frac{q}{2} + cx \right) \left(\frac{b}{2} + \frac{q}{2} + cx \right)$

■

Rule 1.2.1.1.2.1.1: If $b^2 - 4ac \neq 0 \wedge p \in \mathbb{Z}^+ \wedge \text{PerfectSquare}[b^2 - 4ac]$, let $q = \sqrt{b^2 - 4ac}$, then

$$\int (a + bx + cx^2)^p dx \rightarrow \frac{1}{c^p} \int \left(\frac{b}{2} - \frac{q}{2} + cx \right)^p \left(\frac{b}{2} + \frac{q}{2} + cx \right)^p dx$$

Program code:

```
Int[(a_+b_.**x_+c_.**x_^2)^p_,x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    1/c^p*Int[Simp[b/2-q/2+c*x,x]^p*Simp[b/2+q/2+c*x,x]^p,x] /;
  FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && PerfectSquareQ[b^2-4*a*c]
```

$$2: \int (a + bx + cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge p \in \mathbb{Z}^+ \wedge \neg \text{PerfectSquare}[b^2 - 4ac]$$

Derivation: Algebraic expansion

Rule 1.2.1.1.2.1.2: If $b^2 - 4ac \neq 0 \wedge p \in \mathbb{Z}^+ \wedge \neg \text{PerfectSquare}[b^2 - 4ac]$, then

$$\int (a + bx + cx^2)^p dx \rightarrow \int \text{ExpandIntegrand}[(a + bx + cx^2)^p, x] dx$$

Program code:

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && Not[PerfectSquareQ[b^2-4*a*c]]
```

$$2: \int (a + bx + cx^2)^p dx \text{ when } b^2 - 4ac \neq 0 \wedge p > 0 \wedge p \notin \mathbb{Z}$$

Reference: G&R 2.260.2, CRC 245, A&S 3.3.37

Derivation: Quadratic recurrence 1b with $m = -1$, $A = d$ and $B = e$

Rule 1.2.1.1.2.2: If $b^2 - 4ac \neq 0 \wedge p > 0 \wedge p \notin \mathbb{Z}$, then

$$\int (a + bx + cx^2)^p dx \rightarrow \frac{(b + 2cx)(a + bx + cx^2)^p}{2c(2p+1)} - \frac{p(b^2 - 4ac)}{2c(2p+1)} \int (a + bx + cx^2)^{p-1} dx$$

Program code:

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^p/(2*c*(2*p+1)) -
  p*(b^2-4*a*c)/(2*c*(2*p+1))*Int[(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && IntegerQ[4*p]
```

3. $\int (a + bx + cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge 4p \in \mathbb{Z} \wedge p < -1$

1: $\int \frac{1}{(a + bx + cx^2)^{3/2}} dx$ when $b^2 - 4ac \neq 0$

Reference: G&R 2.264.5, CRC 239

Derivation: Quadratic recurrence 2a with $m = 0$, $A = 1$, $B = 0$ and $p = -\frac{3}{2}$

Rule 1.2.1.1.3.1: If $b^2 - 4ac \neq 0$, then

$$\int \frac{1}{(a + bx + cx^2)^{3/2}} dx \rightarrow -\frac{2(b + 2cx)}{(b^2 - 4ac) \sqrt{a + bx + cx^2}}$$

Program code:

```
Int[1/(a_.+b_.*x_+c_.*x_^2)^(3/2),x_Symbol] :=
  -2*(b+2*c*x)/((b^2-4*a*c)*Sqrt[a+b*x+c*x^2]) /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

2: $\int (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge p < -1 \wedge p \neq -\frac{3}{2}$

Reference: G&R 2.171.3, G&R 2.263.3, CRC 113, CRC 241

Derivation: Quadratic recurrence 2a with $m = 0$, $A = 1$ and $B = 0$

Rule 1.2.1.1.3.2: If $b^2 - 4ac \neq 0 \wedge p < -1 \wedge p \neq -\frac{3}{2}$, then

$$\int (a+bx+cx^2)^p dx \rightarrow \frac{(b+2cx)(a+bx+cx^2)^{p+1}}{(p+1)(b^2-4ac)} - \frac{2c(2p+3)}{(p+1)(b^2-4ac)} \int (a+bx+cx^2)^{p+1} dx$$

Program code:

```
Int[(a_.+b_.**x_+c_.**x_^2)^p_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) -
  2*c*(2*p+3)/((p+1)*(b^2-4*a*c))*Int[(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && NeQ[p,-3/2] && IntegerQ[4*p]
```

4. $\int \frac{1}{a+bx+cx^2} dx$ when $b^2 - 4ac \neq 0$

1: $\int \frac{1}{a+bx+cx^2} dx$ when $b^2 - 4ac \neq 0 \wedge b^2 - 4ac > 0 \wedge \text{PerfectSquare}[b^2 - 4ac]$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let $q \rightarrow \sqrt{b^2 - 4ac}$, then $\frac{1}{a+bz+cz^2} = \frac{c}{q} \frac{1}{\frac{b-q}{2}+cz} - \frac{c}{q} \frac{1}{\frac{b+q}{2}+cz}$

■ Rule 1.2.1.1.4.1: If $b^2 - 4ac \neq 0 \wedge b^2 - 4ac > 0 \wedge \text{PerfectSquare}[b^2 - 4ac]$, let $q \rightarrow \sqrt{b^2 - 4ac}$, then

$$\int \frac{1}{a+bx+cx^2} dx \rightarrow \frac{c}{q} \int \frac{1}{\frac{b}{2} - \frac{a}{2} + cx} dx - \frac{c}{q} \int \frac{1}{\frac{b}{2} + \frac{a}{2} + cx} dx$$

Program code:

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    c/q*Int[1/Simp[b/2-q/2+c*x,x],x] - c/q*Int[1/Simp[b/2+q/2+c*x,x],x] /;
  FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && PosQ[b^2-4*a*c] && PerfectSquareQ[b^2-4*a*c]
```

2: $\int \frac{1}{a+bx+cx^2} dx$ when $b^2 - 4ac \notin \mathbb{R} \wedge \frac{b^2-4ac}{b^2} \in \mathbb{R}$

Reference: G&R 2.172.4, CRC 109, A&S 3.3.16

Reference: G&R 2.172.2, CRC 110a, A&S 3.3.17

Derivation: Integration by substitution

Basis: $\frac{1}{a+bx+cx^2} = -\frac{2}{b} \text{Subst}\left[\frac{1}{q-x^2}, x, 1 + \frac{2cx}{b}\right] \partial_x \left(1 + \frac{2cx}{b}\right)$

Rule 1.2.1.1.4.2: If $b^2 - 4ac \notin \mathbb{R}$, let $q \rightarrow \frac{b^2-4ac}{b^2}$, if $q \in \mathbb{R}$, then

$$\int \frac{1}{a+bx+cx^2} dx \rightarrow -\frac{2}{b} \text{Subst}\left[\int \frac{1}{q-x^2} dx, x, 1 + \frac{2cx}{b}\right]$$

Program code:

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  With[{q=1-4*Simplify[a*c/b^2]},
    -2/b*Subst[Int[1/(q-x^2),x],x,1+2*c*x/b] /;
  RationalQ[q] && (EqQ[q^2,1] || Not[RationalQ[b^2-4*a*c]]) /;
  FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

3: $\int \frac{1}{a+bx+cx^2} dx$ when $b^2 - 4ac \neq 0$

Reference: G&R 2.172.2, CRC 110a, A&S 3.3.17

Reference: G&R 2.172.4, CRC 109, A&S 3.3.16

Derivation: Integration by substitution

Basis: $\frac{1}{a+bx+cx^2} = -2 \operatorname{Subst}\left[\frac{1}{b^2-4ac-x^2}, x, b+2cx\right] \partial_x(b+2cx)$

Rule 1.2.1.1.4.3: If $b^2 - 4ac \neq 0$, then

$$\int \frac{1}{a+bx+cx^2} dx \rightarrow -2 \operatorname{Subst}\left[\int \frac{1}{b^2-4ac-x^2} dx, x, b+2cx\right]$$

Program code:

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  -2*Subst[Int[1/Simp[b^2-4*a*c-x^2,x],x],x,b+2*c*x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```


5: $\int (a + bx + cx^2)^p dx$ when $4a - \frac{b^2}{c} > 0$

Derivation: Integration by substitution

Basis: If $4a - \frac{b^2}{c} > 0$, then $(a + bx + cx^2)^p = \frac{1}{2c \left(-\frac{4c}{b^2-4ac}\right)^p} \text{Subst}\left[\left(1 - \frac{x^2}{b^2-4ac}\right)^p, x, b+2cx\right] \partial_x (b+2cx)$

Rule 1.2.1.1.5: If $4a - \frac{b^2}{c} > 0$, then

$$\int (a + bx + cx^2)^p dx \rightarrow \frac{1}{2c \left(-\frac{4c}{b^2-4ac}\right)^p} \text{Subst}\left[\int \left(1 - \frac{x^2}{b^2-4ac}\right)^p dx, x, b+2cx\right]$$

Program code:

```
Int[(a_.+b_.**x_+c_.**x_^2)^p_,x_Symbol] :=
  1/(2*c*(-4*c/(b^2-4*a*c))^p)*Subst[Int[Simp[1-x^2/(b^2-4*a*c),x]^p,x],x,b+2*c*x] /;
FreeQ[{a,b,c,p},x] && GtQ[4*a-b^2/c,0]
```

6. $\int \frac{1}{\sqrt{a+bx+cx^2}} dx$ when $b^2 - 4ac \neq 0$

1: $\int \frac{1}{\sqrt{bx+cx^2}} dx$

Derivation: Integration by substitution

Basis: $\frac{1}{\sqrt{bx+cx^2}} = 2 \operatorname{Subst}\left[\frac{1}{1-cx^2}, x, \frac{x}{\sqrt{bx+cx^2}}\right] \partial_x \frac{x}{\sqrt{bx+cx^2}}$

Rule 1.2.1.1.6.1:

$$\int \frac{1}{\sqrt{bx+cx^2}} dx \rightarrow 2 \operatorname{Subst}\left[\int \frac{1}{1-cx^2} dx, x, \frac{x}{\sqrt{bx+cx^2}}\right]$$

Program code:

```
Int[1/Sqrt[b_.*x_+c_.*x_^2],x_Symbol] :=
  2*Subst[Int[1/(1-c*x^2),x],x,x/Sqrt[b*x+c*x^2]] /;
FreeQ[{b,c},x]
```

2: $\int \frac{1}{\sqrt{a+bx+cx^2}} dx$ when $b^2 - 4ac \neq 0$

Reference: G&R 2.261.1, CRC 237a, A&S 3.3.33

Reference: CRC 238

Derivation: Integration by substitution

Basis: $\frac{1}{\sqrt{a+bx+cx^2}} = 2 \operatorname{Subst}\left[\frac{1}{4c-x^2}, x, \frac{b+2cx}{\sqrt{a+bx+cx^2}}\right] \partial_x \frac{b+2cx}{\sqrt{a+bx+cx^2}}$

Rule 1.2.1.1.6.2: If $b^2 - 4ac \neq 0$, then

$$\int \frac{1}{\sqrt{a+bx+cx^2}} dx \rightarrow 2 \text{ Subst} \left[\int \frac{1}{4c-x^2} dx, x, \frac{b+2cx}{\sqrt{a+bx+cx^2}} \right]$$

Program code:

```
Int[1/Sqrt[a+_.*x+_.*x^2],x_Symbol] :=
  2*Subst[Int[1/(4*c-x^2),x],x,(b+2*c*x)/Sqrt[a+b*x+c*x^2]] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

7. $\int (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge 3 \leq \text{Denominator}[p] \leq 4$

1: $\int (bx+cx^2)^p dx$ when $3 \leq \text{Denominator}[p] \leq 4$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(bx+cx^2)^p}{\left(-\frac{c(bx+cx^2)}{b^2}\right)^p} = 0$

Note: If this optional rule is deleted, the resulting antiderivative is less compact but real when the integrand is real.

Rule 1.2.1.1.7.1: If $3 \leq \text{Denominator}[p] \leq 4$, then

$$\int (bx+cx^2)^p dx \rightarrow \frac{(bx+cx^2)^p}{\left(-\frac{c(bx+cx^2)}{b^2}\right)^p} \int \left(-\frac{cx}{b} - \frac{c^2 x^2}{b^2}\right)^p dx$$

Program code:

```
Int[(b_*x+c_*x^2)^p_,x_Symbol] :=
  (b*x+c*x^2)^p/(-c*(b*x+c*x^2)/(b^2))^p*Int[(-c*x/b-c^2*x^2/b^2)^p,x] /;
FreeQ[{b,c},x] && RationalQ[p] && 3<Denominator[p]<=4
```

x: $\int (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge 3 \leq \text{Denominator}[p] \leq 4$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+bx+cx^2)^p}{\left(-\frac{c(a+bx+cx^2)}{b^2-4ac}\right)^p} = 0$

Rule 1.2.1.1.7.2: If $b^2 - 4ac \neq 0 \wedge 3 \leq \text{Denominator}[p] \leq 4$, then

$$\int (a+bx+cx^2)^p dx \rightarrow \frac{(a+bx+cx^2)^p}{\left(-\frac{c(a+bx+cx^2)}{b^2-4ac}\right)^p} \int \left(-\frac{ac}{b^2-4ac} - \frac{bcx}{b^2-4ac} - \frac{c^2x^2}{b^2-4ac}\right)^p dx$$

Program code:

```
(* Int[(a_.+b_.*x+c_.*x^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^p/(-c*(a+b*x+c*x^2)/(b^2-4*a*c))^p*Int[(-a*c/(b^2-4*a*c)-b*c*x/(b^2-4*a*c)-c^2*x^2/(b^2-4*a*c))^p,x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && RationalQ[p] && 3<Denominator[p]<=4 *)
```

2: $\int (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge 3 \leq \text{Denominator}[p] \leq 4$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $d \in \mathbb{Z}^+$, then $(a+bx+cx^2)^p = \frac{d\sqrt{(b+2cx)^2}}{b+2cx} \text{Subst}\left[\frac{x^{d(p+1)-1}}{\sqrt{b^2-4ac+4cx^d}}, x, (a+bx+cx^2)^{1/d}\right] \partial_x (a+bx+cx^2)^{1/d}$

Basis: $\partial_x \frac{\sqrt{(b+2cx)^2}}{b+2cx} = 0$

Note: Since $d \leq 4$, resulting integrand is an elliptic integral.

Rule 1.2.1.1.7.2: If $b^2 - 4ac \neq 0$, let $d \rightarrow \text{Denominator}[p]$, if $3 \leq d \leq 4$, then

$$\int (a+bx+cx^2)^p dx \rightarrow \frac{d\sqrt{(b+2cx)^2}}{b+2cx} \text{Subst}\left[\int \frac{x^{d(p+1)-1}}{\sqrt{b^2-4ac+4cx^d}} dx, x, (a+bx+cx^2)^{1/d}\right]$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{d=Denominator[p]},
    d*Sqrt[(b+2*c*x)^2]/(b+2*c*x)*Subst[Int[x^(d*(p+1)-1)/Sqrt[b^2-4*a*c+4*c*x^d],x],x,(a+b*x+c*x^2)^(1/d)] /;
    3<=d<=4] /;
  FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && RationalQ[p]
```

H: $\int (a+bx+cx^2)^p dx$ when $b^2-4ac \neq 0 \wedge 4p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: Let $q = \sqrt{b^2-4ac}$, then $\partial_x \frac{(a+bx+cx^2)^p}{(b+q+2cx)^p (b-q+2cx)^p} = 0$

■

Rule 1.2.1.1.H: If $b^2-4ac \neq 0 \wedge 4p \notin \mathbb{Z}$, let $q = \sqrt{b^2-4ac}$, then

$$\begin{aligned} \int (a+bx+cx^2)^p dx &\rightarrow \frac{(a+bx+cx^2)^p}{(b+q+2cx)^p (b-q+2cx)^p} \int (b+q+2cx)^p (b-q+2cx)^p dx \\ &\rightarrow -\frac{(a+bx+cx^2)^{p+1}}{q(p+1) \left(\frac{q-b-2cx}{2q}\right)^{p+1}} \text{Hypergeometric2F1}\left[-p, p+1, p+2, \frac{b+q+2cx}{2q}\right] \end{aligned}$$

Program code:

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    -(a+b*x+c*x^2)^(p+1)/(q*(p+1)*((q-b-2*c*x)/(2*q))^(p+1))*Hypergeometric2F1[-p,p+1,p+2,(b+q+2*c*x)/(2*q)] /;
    FreeQ[{a,b,c,p},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[4*p]]
```

S: $\int (a + bu + cu^2)^p dx$ when $u = d + ex$

Derivation: Integration by substitution

Rule 1.2.1.1.S: If $u = d + ex$, then

$$\int (a + bu + cu^2)^p dx \rightarrow \frac{1}{e} \text{Subst} \left[\int (a + bx + cx^2)^p dx, x, u \right]$$

Program code:

```
Int[(a_.+b_.*u_.+c_.*u_^2)^p_,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x+c*x^2)^p,x],x,u] /;
FreeQ[{a,b,c,p},x] && LinearQ[u,x] && NeQ[u,x]
```