Rules for integrands of the form $(a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,\left(A + B\,\text{Sec}[e + f\,x] + C\,\text{Sec}[e + f\,x]^2\right)$

1. $\int (a + b\,\text{Sec}[e + f\,x])\,(d\,\text{Sec}[e + f\,x])^n\,\left(A + B\,\text{Sec}[e + f\,x] + C\,\text{Sec}[e + f\,x]^2\right)\,dx$

   1: $\int (a + b\,\text{Sec}[e + f\,x])\,(d\,\text{Sec}[e + f\,x])^n\,\left(A + B\,\text{Sec}[e + f\,x] + C\,\text{Sec}[e + f\,x]^2\right)\,dx$ when $n < -1$

Derivation: Algebraic expansion, nondegenerate secant recurrence 1c with $c \to 1,\ d \to 0,\ A \to c,\ B \to d,\ C \to 0,\ n \to 0,\ p \to 0$ and algebraic simplification

Basis: $A + B\,z + C\,z^2 == A + \frac{(d\,z)\,(B+C\,z)}{d}$

Rule: If $n < -1$, then

$$\int (a + b\,\text{Sec}[e + f\,x])\,(d\,\text{Sec}[e + f\,x])^n\,\left(A + B\,\text{Sec}[e + f\,x] + C\,\text{Sec}[e + f\,x]^2\right)\,dx \longrightarrow$$

$$A\int (a + b\,\text{Sec}[e + f\,x])\,(d\,\text{Sec}[e + f\,x])^n\,dx + \frac{1}{d}\int (a + b\,\text{Sec}[e + f\,x])\,(d\,\text{Sec}[e + f\,x])^{n+1}\,(B + C\,\text{Sec}[e + f\,x])\,dx \longrightarrow$$

$$-\frac{A\,a\,\text{Tan}[e + f\,x]\,(d\,\text{Sec}[e + f\,x])^n}{f\,n} + \frac{1}{d\,n}\int (d\,\text{Sec}[e + f\,x])^{n+1}\,\left(n\,(B\,a + A\,b) + (n\,(a\,C + B\,b) + A\,a\,(n + 1))\,\text{Sec}[e + f\,x] + b\,C\,n\,\text{Sec}[e + f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*a*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*n) +
  1/(d*n)*Int[(d*Csc[e+f*x])^(n+1)*Simp[n*(B*a+A*b)+(n*(a*C+B*b)+A*a*(n+1))*Csc[e+f*x]+b*C*n*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C},x] && LtQ[n,-1]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*a*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*n) +
  1/(d*n)*Int[(d*Csc[e+f*x])^(n+1)*Simp[A*b*n+a*(C*n+A*(n+1))*Csc[e+f*x]+b*C*n*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C},x] && LtQ[n,-1]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $n \nless -1$

Derivation: Algebraic expansion, nondegenerate secant recurrence 1b with $c\to 0,\ d\to 1,\ A\to a\,c,\ B\to b\,c+a\,d,\ C\to b\,d,\ m\to m+1,\ n\to 0,\ p\to 0$ and algebraic simplification

Basis: $A+B\,z+C\,z^2 == \frac{C\,(d\,z)^2}{d^2}+A+B\,z$

Rule: If $n \nless -1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx\ \longrightarrow$$

$$\frac{C}{d^2}\int (a+b\,\text{Sec}[e+f\,x])\,(d\,\text{Sec}[e+f\,x])^{n+2}\,dx+\int (a+b\,\text{Sec}[e+f\,x])\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\ \longrightarrow$$

$$\frac{b\,C\,\text{Sec}[e+f\,x]\,\text{Tan}[e+f\,x]\,(d\,\text{Sec}[e+f\,x])^n}{f\,(n+2)}+$$

$$\frac{1}{n+2}\int (d\,\text{Sec}[e+f\,x])^n\,\left(A\,a\,(n+2)+(B\,a\,(n+2)+b\,(C\,(n+1)+A\,(n+2)))\,\text{Sec}[e+f\,x]+(a\,C+B\,b)\,(n+2)\,\text{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_.*(a_+b_.*csc[e_.+f_.*x_])*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -b*C*Csc[e+f*x]*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(n+2)) +
  1/(n+2)*Int[(d*Csc[e+f*x])^n*Simp[A*a*(n+2)+(B*a*(n+2)+b*(C*(n+1)+A*(n+2)))*Csc[e+f*x]+(a*C+B*b)*(n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,n},x] && Not[LtQ[n,-1]]
```

```
Int[(d_.*csc[e_.+f_.*x_])^n_.*(a_+b_.*csc[e_.+f_.*x_])*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -b*C*Csc[e+f*x]*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(n+2)) +
  1/(n+2)*Int[(d*Csc[e+f*x])^n*Simp[A*a*(n+2)+b*(C*(n+1)+A*(n+2))*Csc[e+f*x]+a*C*(n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C,n},x] && Not[LtQ[n,-1]]
```

2. $\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx$

1. $\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx$ when $m<-1$

1: $\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx$ when $m<-1\,\land\,a^2-b^2=0$

Derivation: Algebraic expansion, singly degenerate secant recurrence 2b with $A\to 1,\ B\to 0,\ n\to 1,\ p\to 0$ and algebraic simplification

Basis: If $a^2-b^2=0$, then $A+B\,z+C\,z^2=\frac{a\,A-b\,B+a\,C}{a}+\frac{(a+b\,z)\,(b\,B-a\,C+b\,C\,z)}{b^2}$

Rule: If $m<-1\,\land\,a^2-b^2=0$, then

$$\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx\,\longrightarrow$$

$$\frac{a\,A-b\,B+a\,C}{a}\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,dx+\frac{1}{b^2}\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(b\,B-a\,C+b\,C\,\mathrm{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(a\,A-b\,B+a\,C)\,\mathrm{Tan}[e+f\,x]\,\mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m}{a\,f\,(2\,m+1)}-$$

$$\frac{1}{a\,b\,(2\,m+1)}\int \mathrm{Sec}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(a\,B-b\,C-2\,A\,b\,(m+1)-(b\,B\,(m+2)-a\,(A\,(m+2)-C\,(m-1)))\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -(a*A-b*B+a*C)*Cot[e+f*x]*Csc[e+f*x]*(a+b*Csc[e+f*x])^m/(a*f*(2*m+1)) -
  1/(a*b*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[a*B-b*C-2*A*b*(m+1)-(b*B*(m+2)-a*(A*(m+2)-C*(m-1)))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && LtQ[m,-1] && EqQ[a^2-b^2,0]
```

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -(A+C)*Cot[e+f*x]*Csc[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(2*m+1)) -
  1/(a*b*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[-b*C-2*A*b*(m+1)+a*(A*(m+2)-C*(m-1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && LtQ[m,-1] && EqQ[a^2-b^2,0]
```

2: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,\left(A+B\,\text{Sec}[e+f x]+C\,\text{Sec}[e+f x]^2\right)dx$ when $m<-1\land a^2-b^2\neq 0$

Derivation: Secant recurrence 2a with $n\to 1$

Rule: If $m<-1\land a^2-b^2\neq 0$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,\left(A+B\,\text{Sec}[e+f x]+C\,\text{Sec}[e+f x]^2\right)dx\longrightarrow$$
$$\frac{\left(A\,b^2-a\,b\,B+a^2\,C\right)\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}}{b\,f\,(m+1)\,\left(a^2-b^2\right)}+$$
$$\frac{1}{b\,(m+1)\,\left(a^2-b^2\right)}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}\cdot$$
$$\left(b\,(a\,A-b\,B+a\,C)\,(m+1)-\left(A\,b^2-a\,b\,B+a^2\,C+b\,(A\,b-a\,B+b\,C)\,(m+1)\right)\text{Sec}[e+f x]\right)dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -(A*b^2-a*b*B+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) +
  1/(b*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[b*(a*A-b*B+a*C)*(m+1)-(A*b^2-a*b*B+a^2*C+b*(A*b-a*B+b*C)*(m+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && LtQ[m,-1] && NeQ[a^2-b^2,0]
```

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -(A*b^2+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) +
  1/(b*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[a*b*(A+C)*(m+1)-(A*b^2+a^2*C+b*(A*b+b*C)*(m+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && LtQ[m,-1] && NeQ[a^2-b^2,0]
```

2: $\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x]+C\,\text{Sec}[e+f x]^2)\,dx$ when $m \not< -1$

Derivation: Secant recurrence 3a with $n \to 1$

Rule: If $m \not< -1$, then

$$\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(A+B\,\text{Sec}[e+f x]+C\,\text{Sec}[e+f x]^2)\,dx \longrightarrow$$
$$\frac{C\,\text{Tan}[e+f x]\,(a+b\,\text{Sec}[e+f x])^{m+1}}{b\,f\,(m+2)}+$$
$$\frac{1}{b\,(m+2)}\int \text{Sec}[e+f x]\,(a+b\,\text{Sec}[e+f x])^m\,(b\,A\,(m+2)+b\,C\,(m+1)+(b\,B\,(m+2)-a\,C)\,\text{Sec}[e+f x])\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*Simp[b*A*(m+2)+b*C*(m+1)+(b*B*(m+2)-a*C)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && Not[LtQ[m,-1]]
```

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*Simp[b*A*(m+2)+b*C*(m+1)-a*C*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && Not[LtQ[m,-1]]
```

3 $\int (a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,(A + B\,\text{Sec}[e + f\,x] + C\,\text{Sec}[e + f\,x]^2)\,dx$ when $a^2 - b^2 == 0$

1: $\int (a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,(A + B\,\text{Sec}[e + f\,x] + C\,\text{Sec}[e + f\,x]^2)\,dx$ when $a^2 - b^2 == 0 \land m < -\frac{1}{2}$

Derivation: Algebraic expansion, singly degenerate secant recurrence 2b with $A \to 1,\ B \to 0,\ p \to 0$ and algebraic simplification

Basis: If $a^2 - b^2 == 0$, then $A + B\,z + C\,z^2 == \frac{a\,A - b\,B + a\,C}{a} + \frac{(a + b\,z)\,(b\,B - a\,C + b\,C\,z)}{b^2}$

Rule: If $a^2 - b^2 == 0 \land m < -\frac{1}{2}$, then

$$\int (a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,(A + B\,\text{Sec}[e + f\,x] + C\,\text{Sec}[e + f\,x]^2)\,dx \longrightarrow$$

$$\frac{a\,A - b\,B + a\,C}{a}\int (a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n\,dx + \frac{1}{b^2}\int (a + b\,\text{Sec}[e + f\,x])^{m+1}\,(d\,\text{Sec}[e + f\,x])^n\,(b\,B - a\,C + b\,C\,\text{Sec}[e + f\,x])\,dx \longrightarrow$$

$$\frac{(a\,A - b\,B + a\,C)\,\text{Tan}[e + f\,x]\,(a + b\,\text{Sec}[e + f\,x])^m\,(d\,\text{Sec}[e + f\,x])^n}{a\,f\,(2\,m + 1)} -$$
$$\frac{1}{a\,b\,(2\,m + 1)}\int (a + b\,\text{Sec}[e + f\,x])^{m+1}\,(d\,\text{Sec}[e + f\,x])^n\,\cdot$$
$$(a\,B\,n - b\,C\,n - A\,b\,(2\,m + n + 1) - (b\,B\,(m + n + 1) - a\,(A\,(m + n + 1) - C\,(m - n)))\,\text{Sec}[e + f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -(a*A-b*B+a*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(a*f*(2*m+1)) -
  1/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[a*B*n-b*C*n-A*b*(2*m+n+1)-(b*B*(m+n+1)-a*(A*(m+n+1)-C*(m-n)))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,n},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -a*(A+C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(a*f*(2*m+1)) +
  1/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[b*C*n+A*b*(2*m+n+1)-(a*(A*(m+n+1)-C*(m-n)))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,C,n},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

2. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2=0\,\wedge\,m\not<-\frac{1}{2}$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2=0\,\wedge\,m\not<-\frac{1}{2}\,\wedge\,\left(n<-\frac{1}{2}\,\vee\,m+n+1=0\right)$

Derivation: Algebraic expansion and singly degenerate secant recurrence 1c with $A\to 1,\ B\to 0,\ p\to 0$

Basis: $A+B\,z+C\,z^2=A+\frac{(d\,z)\,(B+C\,z)}{d}$

Rule: If $a^2-b^2=0\,\wedge\,m\not<-\frac{1}{2}\,\wedge\,\left(n<-\frac{1}{2}\,\vee\,m+n+1=0\right)$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx\,\longrightarrow$$

$$A\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,dx+\frac{1}{d}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n+1}\,(B+C\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{A\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n}{f\,n}-\frac{1}{b\,d\,n}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n+1}\,(a\,A\,m-b\,B\,n-b\,(A\,(m+n+1)+C\,n)\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(b*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*Simp[a*A*m-b*B*n-b*(A*(m+n+1)+C*n)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,m},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]] && (LtQ[n,-1/2] || EqQ[m+n+1,0])
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(b*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*Simp[a*A*m-b*(A*(m+n+1)+C*n)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,C,m},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]] && (LtQ[n,-1/2] || EqQ[m+n+1,0])
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,\left(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2\right)\,dx$ when $a^2-b^2==0\,\land\,m\not<-\frac{1}{2}\,\land\,n\not<-\frac{1}{2}\,\land\,m+n+1\neq 0$

Derivation: Nondegenerate secant recurrence 1b with $p\to 0$ and $a^2-b^2==0$

Derivation: Algebraic expansion and singly degenerate secant recurrence 2c with $A\to c,\ B\to d,\ n\to n+1,\ p\to 0$

Basis: $A+B\,z+C\,z^2==\frac{C\,(d\,z)^2}{d^2}+A+B\,z$

Rule: If $a^2-b^2==0\,\land\,m\not<-\frac{1}{2}\,\land\,m+n+1\neq 0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,\left(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2\right)\,dx\,\longrightarrow$$

$$\frac{C}{d^2}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^{n+2}\,dx+\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{C\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n}{f\,(m+n+1)}+$$
$$\frac{1}{b\,(m+n+1)}\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,\left(A\,b\,(m+n+1)+b\,C\,n+(a\,C\,m+b\,B\,(m+n+1))\,\text{Sec}[e+f\,x]\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*(m+n+1)) +
  1/(b*(m+n+1))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n*Simp[A*b*(m+n+1)+b*C*n+(a*C*m+b*B*(m+n+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,m,n},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]] && Not[LtQ[n,-1/2]] && NeQ[m+n+1,0]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*(m+n+1)) +
  1/(b*(m+n+1))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n*Simp[A*b*(m+n+1)+b*C*n+a*C*m*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,C,m,n},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]] && Not[LtQ[n,-1/2]] && NeQ[m+n+1,0]
```

4. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(d\,\text{Sec}[e+f\,x])^n\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0$

1. $\int \text{Sec}[e+f\,x]^2\,(a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0$

1: $\int \text{Sec}[e+f\,x]^2\,(a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0\,\wedge\,m<-1$

Derivation: Algebraic expansion, nondegenerate secant recurrence 1c with $c\to 1,\ d\to 0,\ A\to c,\ B\to d,\ C\to 0,\ n\to 0,\ p\to 0$ and algebraic simplification

Basis: $A+B\,z+C\,z^2 == \frac{A\,b^2-a\,b\,B+a^2\,C}{b^2}+\frac{(a+b\,z)\,(b\,B-a\,C+b\,C\,z)}{b^2}$

Rule: If $a^2-b^2\neq 0\,\wedge\,m<-1$, then

$$\int \text{Sec}[e+f\,x]^2\,(a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx\,\longrightarrow$$

$$\frac{A\,b^2-a\,b\,B+a^2\,C}{b^2}\int \text{Sec}[e+f\,x]^2\,(a+b\,\text{Sec}[e+f\,x])^m\,dx+\frac{1}{b^2}\int \text{Sec}[e+f\,x]^2\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\,(b\,B-a\,C+b\,C\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{a\,(A\,b^2-a\,b\,B+a^2\,C)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}}{b^2\,f\,(m+1)\,(a^2-b^2)}-$$

$$\frac{1}{b^2\,(m+1)\,(a^2-b^2)}\int \text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}\,\cdot$$

$$(b\,(m+1)\,(-a\,(b\,B-a\,C)+A\,b^2)+(b\,B\,(a^2+b^2\,(m+1))-a\,(A\,b^2\,(m+2)+C\,(a^2+b^2\,(m+1))))\,\text{Sec}[e+f\,x]-b\,C\,(m+1)\,(a^2-b^2)\,\text{Sec}[e+f\,x]^2)\,dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  a*(A*b^2-a*b*B+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b^2*f*(m+1)*(a^2-b^2)) -
  1/(b^2*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[b*(m+1)*(-a*(b*B-a*C)+A*b^2)+
      (b*B*(a^2+b^2*(m+1))-a*(A*b^2*(m+2)+C*(a^2+b^2*(m+1))))*Csc[e+f*x]-
      b*C*(m+1)*(a^2-b^2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  a*(A*b^2+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b^2*f*(m+1)*(a^2-b^2)) -
  1/(b^2*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[b*(m+1)*(a^2*C+A*b^2)-a*(A*b^2*(m+2)+C*(a^2+b^2*(m+1)))*Csc[e+f*x]-b*C*(m+1)*(a^2-b^2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,C},x] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

2: $\int \text{Sec}[e+f x]^2 (a+b \text{Sec}[e+f x])^m (A+B \text{Sec}[e+f x]+C \text{Sec}[e+f x]^2) \, dx$ when $a^2-b^2 \neq 0 \land m \nless -1$

Derivation: Algebraic expansion, nondegenerate secant recurrence 1b with $c \to 0$, $d \to 1$, $A \to a c$, $B \to b c + a d$, $C \to b d$, $m \to m+1$, $n \to 0$, $p \to 0$ and algebraic simplification

Basis: $A + B z + C z^2 == \frac{C (a+b z)^2}{b^2} + \frac{A b^2 - a^2 C + b (b B - 2 a C) z}{b^2}$

Rule: If $a^2 - b^2 \neq 0 \land m \nless -1$, then

$$\int \text{Sec}[e+f x]^2 (a+b \text{Sec}[e+f x])^m (A+B \text{Sec}[e+f x]+C \text{Sec}[e+f x]^2) \, dx \longrightarrow$$

$$\frac{C}{b^2} \int \text{Sec}[e+f x]^2 (a+b \text{Sec}[e+f x])^{m+2} \, dx + \frac{1}{b^2} \int \text{Sec}[e+f x]^2 (a+b \text{Sec}[e+f x])^m \left(A b^2 - a^2 C + b (b B - 2 a C) \text{Sec}[e+f x]\right) dx \longrightarrow$$

$$\frac{C \text{Sec}[e+f x] \text{Tan}[e+f x] (a+b \text{Sec}[e+f x])^{m+1}}{b f (m+3)} +$$
$$\frac{1}{b (m+3)} \int \text{Sec}[e+f x] (a+b \text{Sec}[e+f x])^m \left(a C + b (C (m+2) + A (m+3)) \text{Sec}[e+f x] - (2 a C - b B (m+3)) \text{Sec}[e+f x]^2\right) dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Csc[e+f*x]*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+3)) +
  1/(b*(m+3))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*
    Simp[a*C+b*(C*(m+2)+A*(m+3))*Csc[e+f*x]-(2*a*C-b*B*(m+3))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]]
```

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Csc[e+f*x]*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+3)) +
  1/(b*(m+3))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*Simp[a*C+b*(C*(m+2)+A*(m+3))*Csc[e+f*x]-2*a*C*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]]
```

2. $\int (a + b\,\mathrm{Sec}[e + f\,x])^m\, (d\,\mathrm{Sec}[e + f\,x])^n\, (A + B\,\mathrm{Sec}[e + f\,x] + C\,\mathrm{Sec}[e + f\,x]^2)\, dx$ when $a^2 - b^2 \neq 0 \wedge m > 0$

1: $\int (a + b\,\mathrm{Sec}[e + f\,x])^m\, (d\,\mathrm{Sec}[e + f\,x])^n\, (A + B\,\mathrm{Sec}[e + f\,x] + C\,\mathrm{Sec}[e + f\,x]^2)\, dx$ when $a^2 - b^2 \neq 0 \wedge m > 0 \wedge n \leq -1$

Derivation: Nondegenerate secant recurrence 1a with $p \to 0$

Rule: If $a^2 - b^2 \neq 0 \wedge m > 0 \wedge n \leq -1$, then

$$\int (a + b\,\mathrm{Sec}[e + f\,x])^m\, (d\,\mathrm{Sec}[e + f\,x])^n\, (A + B\,\mathrm{Sec}[e + f\,x] + C\,\mathrm{Sec}[e + f\,x]^2)\, dx \to$$

$$-\frac{A\,\mathrm{Tan}[e + f\,x]\,(a + b\,\mathrm{Sec}[e + f\,x])^m\,(d\,\mathrm{Sec}[e + f\,x])^n}{f\,n} -$$

$$\frac{1}{d\,n}\int (a + b\,\mathrm{Sec}[e + f\,x])^{m-1}\,(d\,\mathrm{Sec}[e + f\,x])^{n+1}\,\left(A\,b\,m - a\,B\,n - (b\,B\,n + a\,(C\,n + A\,(n + 1)))\,\mathrm{Sec}[e + f\,x] - b\,(C\,n + A\,(m + n + 1))\,\mathrm{Sec}[e + f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(d*n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n+1)*
    Simp[A*b*m-a*B*n-(b*B*n+a*(C*n+A*(n+1)))*Csc[e+f*x]-b*(C*n+A*(m+n+1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C},x] && NeQ[a^2-b^2,0] && GtQ[m,0] && LeQ[n,-1]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(d*n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n+1)*
    Simp[A*b*m-a*(C*n+A*(n+1))*Csc[e+f*x]-b*(C*n+A*(m+n+1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C},x] && NeQ[a^2-b^2,0] && GtQ[m,0] && LeQ[n,-1]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)dx$ when $a^2-b^2\neq 0\,\land\,m>0\,\land\,n\nleq -1$

Derivation: Nondegenerate secant recurrence 1b with $p\to 0$

Rule: If $a^2-b^2\neq 0\,\land\,m>0\,\land\,n\nleq -1$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)dx\,\longrightarrow$$

$$\frac{C\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n}{f\,(m+n+1)}+\frac{1}{m+n+1}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m-1}\,(d\,\mathrm{Sec}[e+f\,x])^n\,\cdot$$
$$\left(a\,A\,(m+n+1)+a\,C\,n+\left((A\,b+a\,B)\,(m+n+1)+b\,C\,(m+n)\right)\mathrm{Sec}[e+f\,x]+(b\,B\,(m+n+1)+a\,C\,m)\,\mathrm{Sec}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*(m+n+1)) +
  1/(m+n+1)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n*
    Simp[a*A*(m+n+1)+a*C*n+((A*b+a*B)*(m+n+1)+b*C*(m+n))*Csc[e+f*x]+(b*B*(m+n+1)+a*C*m)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,n},x] && NeQ[a^2-b^2,0] && GtQ[m,0] && Not[LeQ[n,-1]]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*(m+n+1)) +
  1/(m+n+1)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n*
    Simp[a*A*(m+n+1)+a*C*n+b*(A*(m+n+1)+C*(m+n))*Csc[e+f*x]+a*C*m*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C,n},x] && NeQ[a^2-b^2,0] && GtQ[m,0] && Not[LeQ[n,-1]]
```

3. $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0\wedge m<-1$

1: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0\wedge m<-1\wedge n>0$

Derivation: Nondegenerate secant recurrence 1a with $p\to 0$

Rule: If $a^2-b^2\neq 0\wedge m<-1\wedge n>0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx\,\longrightarrow$$
$$\left(\left(d\,\left(A\,b^2-a\,b\,B+a^2\,C\right)\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}\right)\Big/\left(b\,f\,\left(a^2-b^2\right)\,(m+1)\right)\right)+$$
$$\frac{d}{b\,\left(a^2-b^2\right)\,(m+1)}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}\,\cdot$$
$$\left(A\,b^2\,(n-1)-a\,(b\,B-a\,C)\,(n-1)+b\,(a\,A-b\,B+a\,C)\,(m+1)\,\mathrm{Sec}[e+f\,x]-\left(b\,(A\,b-a\,B)\,(m+n+1)+C\,\left(a^2\,n+b^2\,(m+1)\right)\right)\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -d*(A*b^2-a*b*B+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)/(b*f*(a^2-b^2)*(m+1)) +
  d/(b*(a^2-b^2)*(m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*
    Simp[A*b^2*(n-1)-a*(b*B-a*C)*(n-1)+
      b*(a*A-b*B+a*C)*(m+1)*Csc[e+f*x]-
      (b*(A*b-a*B)*(m+n+1)+C*(a^2*n+b^2*(m+1)))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[n,0]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -d*(A*b^2+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)/(b*f*(a^2-b^2)*(m+1)) +
  d/(b*(a^2-b^2)*(m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*
    Simp[A*b^2*(n-1)+a^2*C*(n-1)+a*b*(A+C)*(m+1)*Csc[e+f*x]-(A*b^2*(m+n+1)+C*(a^2*n+b^2*(m+1)))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[n,0]
```

2: $\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx$ when $a^2 - b^2 \neq 0 \land m < -1 \land n \ngtr 0$

Derivation: Nondegenerate secant recurrence 1c with $p \to 0$

Rule: If $a^2 - b^2 \neq 0 \land m < -1 \land n \ngtr 0$, then

$$\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx \longrightarrow$$

$$-\left(\left(\left(A b^2 - a b B + a^2 C\right) \operatorname{Tan}[e + f x] (a + b \operatorname{Sec}[e + f x])^{m+1} (d \operatorname{Sec}[e + f x])^n\right) / \left(a f (m + 1) \left(a^2 - b^2\right)\right)\right) + \frac{1}{a (m + 1) \left(a^2 - b^2\right)} \int (a + b \operatorname{Sec}[e + f x])^{m+1} (d \operatorname{Sec}[e + f x])^n \cdot$$
$$\left(a (a A - b B + a C) (m + 1) - \left(A b^2 - a b B + a^2 C\right) (m + n + 1) - a (A b - a B + b C) (m + 1) \operatorname{Sec}[e + f x] + \left(A b^2 - a b B + a^2 C\right) (m + n + 2) \operatorname{Sec}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2-a*b*B+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[a*(a*A-b*B+a*C)*(m+1)-(A*b^2-a*b*B+a^2*C)*(m+n+1)-
      a*(A*b-a*B+b*C)*(m+1)*Csc[e+f*x]+
      (A*b^2-a*b*B+a^2*C)*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,n},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && Not[ILtQ[m+1/2,0] && ILtQ[n,0]]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[a^2*(A+C)*(m+1)-(A*b^2+a^2*C)*(m+n+1)-a*b*(A+C)*(m+1)*Csc[e+f*x]+(A*b^2+a^2*C)*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C,n},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && Not[ILtQ[m+1/2,0] && ILtQ[n,0]]
```

4: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$ when $a^2-b^2\neq 0\,\wedge\,n>0$

Derivation: Nondegenerate secant recurrence 1b with $p\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,n>0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)\,dx\,\longrightarrow$$

$$\frac{C\,d\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}}{b\,f\,(m+n+1)}+$$

$$\frac{d}{b\,(m+n+1)}\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n-1}\,\left(a\,C\,(n-1)+(A\,b\,(m+n+1)+b\,C\,(m+n))\,\mathrm{Sec}[e+f\,x]+(b\,B\,(m+n+1)-a\,C\,n)\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)/(b*f*(m+n+1)) +
  d/(b*(m+n+1))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)*
    Simp[a*C*(n-1)+(A*b*(m+n+1)+b*C*(m+n))*Csc[e+f*x]+(b*B*(m+n+1)-a*C*n)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,m},x] && NeQ[a^2-b^2,0] && GtQ[n,0] (* && Not[IGtQ[m,0] && Not[IntegerQ[n]]] *)
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)/(b*f*(m+n+1)) +
  d/(b*(m+n+1))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)*
    Simp[a*C*(n-1)+(A*b*(m+n+1)+b*C*(m+n))*Csc[e+f*x]-a*C*n*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C,m},x] && NeQ[a^2-b^2,0] && GtQ[n,0] (* && Not[IGtQ[m,0] && Not[IntegerQ[n]]] *)
```

5: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$ when $a^2-b^2\neq 0\,\wedge\,n\leq -1$

Derivation: Nondegenerate secant recurrence 1c with $p\to 0$

Rule: If $c^2-d^2\neq 0\,\wedge\,n\leq -1$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^n\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)\,dx\,\longrightarrow$$
$$-\frac{A\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m+1}\,(d\,\mathrm{Sec}[e+f\,x])^n}{a\,f\,n}+$$
$$\frac{1}{a\,d\,n}\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n+1}\,\left(a\,B\,n-A\,b\,(m+n+1)+a\,(A+A\,n+C\,n)\,\mathrm{Sec}[e+f\,x]+A\,b\,(m+n+2)\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*n) +
  1/(a*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*
    Simp[a*B*n-A*b*(m+n+1)+a*(A+A*n+C*n)*Csc[e+f*x]+A*b*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,C,m},x] && NeQ[a^2-b^2,0] && LeQ[n,-1]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*n) +
  1/(a*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*
    Simp[-A*b*(m+n+1)+a*(A+A*n+C*n)*Csc[e+f*x]+A*b*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,C,m},x] && NeQ[a^2-b^2,0] && LeQ[n,-1]
```

6: $\int \frac{A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2}{\sqrt{d\,\mathrm{Sec}[e+f\,x]}\,(a+b\,\mathrm{Sec}[e+f\,x])}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{\sqrt{d z}\ (a+b z)} == \frac{\left(A b^2-a b B+a^2 C\right) (d z)^{3/2}}{a^2 d^2 (a+b z)} + \frac{a A-(A b-a B) z}{a^2 \sqrt{d z}}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{A+B \operatorname{Sec}[e+f x]+C \operatorname{Sec}[e+f x]^2}{\sqrt{d \operatorname{Sec}[e+f x]}\ (a+b \operatorname{Sec}[e+f x])} dx \to \frac{A b^2-a b B+a^2 C}{a^2 d^2} \int \frac{(d \operatorname{Sec}[e+f x])^{3/2}}{a+b \operatorname{Sec}[e+f x]} dx + \frac{1}{a^2} \int \frac{a A-(A b-a B) \operatorname{Sec}[e+f x]}{\sqrt{d \operatorname{Sec}[e+f x]}} dx$$

Program code:

```
Int[(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2)/(Sqrt[d_.*csc[e_.+f_.*x_]]*(a_+b_.*csc[e_.+f_.*x_])),x_Symbol] :=
  (A*b^2-a*b*B+a^2*C)/(a^2*d^2)*Int[(d*Csc[e+f*x])^(3/2)/(a+b*Csc[e+f*x]),x] +
  1/a^2*Int[(a*A-(A*b-a*B)*Csc[e+f*x])/Sqrt[d*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,B,C},x] && NeQ[a^2-b^2,0]
```

```
Int[(A_.+C_.*csc[e_.+f_.*x_]^2)/(Sqrt[d_.*csc[e_.+f_.*x_]]*(a_+b_.*csc[e_.+f_.*x_])),x_Symbol] :=
  (A*b^2+a^2*C)/(a^2*d^2)*Int[(d*Csc[e+f*x])^(3/2)/(a+b*Csc[e+f*x]),x] +
  1/a^2*Int[(a*A-A*b*Csc[e+f*x])/Sqrt[d*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,C},x] && NeQ[a^2-b^2,0]
```

7: $\int \frac{A+B\,Sec[e+f\,x]+C\,Sec[e+f\,x]^2}{\sqrt{d\,Sec[e+f\,x]}\,\sqrt{a+b\,Sec[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z+C\,z^2}{\sqrt{d\,z}}==\frac{C\,(d\,z)^{3/2}}{d^2}+\frac{A+B\,z}{\sqrt{d\,z}}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{A+B\,Sec[e+f\,x]+C\,Sec[e+f\,x]^2}{\sqrt{d\,Sec[e+f\,x]}\,\sqrt{a+b\,Sec[e+f\,x]}}\,dx \to \frac{C}{d^2}\int \frac{(d\,Sec[e+f\,x])^{3/2}}{\sqrt{a+b\,Sec[e+f\,x]}}\,dx+\int \frac{A+B\,Sec[e+f\,x]}{\sqrt{d\,Sec[e+f\,x]}\,\sqrt{a+b\,Sec[e+f\,x]}}\,dx$$

Program code:

```
Int[(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2)/(Sqrt[d_.*csc[e_.+f_.*x_]]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  C/d^2*Int[(d*Csc[e+f*x])^(3/2)/Sqrt[a+b*Csc[e+f*x]],x] +
  Int[(A+B*Csc[e+f*x])/(Sqrt[d*Csc[e+f*x]]*Sqrt[a+b*Csc[e+f*x]]),x] /;
FreeQ[{a,b,d,e,f,A,B,C},x] && NeQ[a^2-b^2,0]
```

```
Int[(A_.+C_.*csc[e_.+f_.*x_]^2)/(Sqrt[d_.*csc[e_.+f_.*x_]]*Sqrt[a_+b_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  C/d^2*Int[(d*Csc[e+f*x])^(3/2)/Sqrt[a+b*Csc[e+f*x]],x] +
  A*Int[1/(Sqrt[d*Csc[e+f*x]]*Sqrt[a+b*Csc[e+f*x]]),x] /;
FreeQ[{a,b,d,e,f,A,C},x] && NeQ[a^2-b^2,0]
```

X: $\int (a+b\,Sec[e+f\,x])^m\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x]+C\,Sec[e+f\,x]^2)\,dx$

Rule:

$$\int (a+b\,Sec[e+f\,x])^m\,(d\,Sec[e+f\,x])^n\,(A+B\,Sec[e+f\,x]+C\,Sec[e+f\,x]^2)\,dx \to$$

$$\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Sec}[e + f x])^n \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(d_.*csc[e_.+f_.*x_])^n_.*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  Unintegrable[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n*(A+B*Csc[e+f*x]+C*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,A,B,C,m,n},x]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(d_.*csc[e_.+f_.*x_])^n_.*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  Unintegrable[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n*(A+C*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,A,C,m,n},x]
```

Rules for integrands of the form $(a + b \operatorname{Sec}[e + f x])^m (c (d \operatorname{Sec}[e + f x])^p)^n \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right)$

1: $\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Cos}[e + f x])^n \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx$ when $n \notin \mathbb{Z} \land m \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $m \in \mathbb{Z}$, then $(a + b \operatorname{Sec}[z])^m \left(A + B \operatorname{Sec}[z] + C \operatorname{Sec}[z]^2\right) == \frac{d^{m+2} (b+a \operatorname{Cos}[z])^m \left(C+B \operatorname{Cos}[z]+A \operatorname{Cos}[z]^2\right)}{(d \operatorname{Cos}[z])^{m+2}}$

Rule: If $n \notin \mathbb{Z} \land m \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sec}[e + f x])^m (d \operatorname{Cos}[e + f x])^n \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx \longrightarrow$$
$$d^{m+2} \int (b + a \operatorname{Cos}[e + f x])^m (d \operatorname{Cos}[e + f x])^{n-m-2} \left(C + B \operatorname{Cos}[e + f x] + A \operatorname{Cos}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*sec[e_.+f_.*x_])^m_.*(d_.*cos[e_.+f_.*x_])^n_*(A_.+B_.*sec[e_.+f_.*x_]+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  d^(m+2)*Int[(b+a*Cos[e+f*x])^m*(d*Cos[e+f*x])^(n-m-2)*(C+B*Cos[e+f*x]+A*Cos[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,A,B,C,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  d^(m+2)*Int[(b+a*Sin[e+f*x])^m*(d*Sin[e+f*x])^(n-m-2)*(C+B*Sin[e+f*x]+A*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,A,B,C,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

```
Int[(a_+b_.*sec[e_.+f_.*x_])^m_.*(d_.*cos[e_.+f_.*x_])^n_*(A_.+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  d^(m+2)*Int[(b+a*Cos[e+f*x])^m*(d*Cos[e+f*x])^(n-m-2)*(C+A*Cos[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,A,C,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(d_.*sin[e_.+f_.*x_])^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  d^(m+2)*Int[(b+a*Sin[e+f*x])^m*(d*Sin[e+f*x])^(n-m-2)*(C+A*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,A,C,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c\,(d\,\mathrm{Sec}[e+f\,x])^p)^n\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,(d\,\mathrm{Sec}[e+f\,x])^p)^n}{(d\,\mathrm{Sec}[e+f\,x])^{n\,p}} == 0$

Rule: If $n\notin\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c\,(d\,\mathrm{Sec}[e+f\,x])^p)^n\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx\;\longrightarrow$$
$$\frac{c^{\mathrm{IntPart}[n]}\,(c\,(d\,\mathrm{Sec}[e+f\,x])^p)^{\mathrm{FracPart}[n]}}{(d\,\mathrm{Sec}[e+f\,x])^{p\,\mathrm{FracPart}[n]}}\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n\,p}\,(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2)\,dx$$

Program code:

```
Int[(a_+b_.*sec[e_.+f_.*x_])^m_.*(c_.*(d_.*sec[e_.+f_.*x_])^p_)^n_*(A_.+B_.*sec[e_.+f_.*x_]+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  c^IntPart[n]*(c*(d*Sec[e+f*x])^p)^FracPart[n]/(d*Sec[e+f*x])^(p*FracPart[n])*
    Int[(a+b*Sec[e+f*x])^m*(d*Sec[e+f*x])^(n*p)*(A+B*Sec[e+f*x]+C*Sec[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n,p},x] && Not[IntegerQ[n]]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_.*(d_.*csc[e_.+f_.*x_])^p_)^n_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  c^IntPart[n]*(c*(d*Csc[e+f*x])^p)^FracPart[n]/(d*Csc[e+f*x])^(p*FracPart[n])*
    Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n*p)*(A+B*Csc[e+f*x]+C*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n,p},x] && Not[IntegerQ[n]]
```

```
Int[(a_+b_.*sec[e_.+f_.*x_])^m_.*(c_.*(d_.*sec[e_.+f_.*x_])^p_)^n_*(A_.+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  c^IntPart[n]*(c*(d*Sec[e+f*x])^p)^FracPart[n]/(d*Sec[e+f*x])^(p*FracPart[n])*
    Int[(a+b*Sec[e+f*x])^m*(d*Sec[e+f*x])^(n*p)*(A+C*Sec[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n,p},x] && Not[IntegerQ[n]]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_.*(d_.*csc[e_.+f_.*x_])^p_)^n_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  c^IntPart[n]*(c*(d*Csc[e+f*x])^p)^FracPart[n]/(d*Csc[e+f*x])^(p*FracPart[n])*
    Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n*p)*(A+C*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n,p},x] && Not[IntegerQ[n]]
```