Rules for integrands of the form $(g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r$

when $b\,c-a\,d\neq 0\ \wedge\ b\,e-a\,f\neq 0\ \wedge\ d\,e-c\,f\neq 0$

0. $\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$

    1. $\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $m\in\mathbb{Z}\ \vee\ g>0$

        1: $\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $(m\in\mathbb{Z}\ \vee\ g>0)\ \wedge\ \frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,(b\,x^n)^p = \frac{1}{b^{\frac{m+1}{n}-1}}\,x^{n-1}\,(b\,x^n)^{p+\frac{m+1}{n}-1}$

Basis: $x^{n-1}\,F[x^n] = \frac{1}{n}\,\text{Subst}[F[x],\,x,\,x^n]\,\partial_x x^n$

Rule 1.1.3.6.0.1.1: If $(m\in\mathbb{Z}\ \vee\ g>0)\ \wedge\ \frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\ \rightarrow\ \frac{g^m}{n\,b^{\frac{m+1}{n}-1}}\,\text{Subst}\Big[\int (b\,x)^{p+\frac{m+1}{n}-1}\,(c+d\,x)^q\,(e+f\,x)^r\,dx,\,x,\,x^n\Big]$$

Program code:

```
Int[(g_.*x_)^m_.*(b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  g^m/(n*b^(Simplify[(m+1)/n]-1))*Subst[Int[(b*x)^(p+Simplify[(m+1)/n]-1)*(c+d*x)^q*(e+f*x)^r,x],x,x^n] /;
FreeQ[{b,c,d,e,f,g,m,n,p,q,r},x] && (IntegerQ[m] || GtQ[g,0]) && IntegerQ[Simplify[(m+1)/n]]
```

        2: $\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $(m\in\mathbb{Z}\ \vee\ g>0)\ \wedge\ \frac{m+1}{n}\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,x^n)^p}{x^{n\,p}} = 0$

Rule 1.1.3.6.0.1.2: If $(m\in\mathbb{Z}\ \vee\ g>0)\ \wedge\ \frac{m+1}{n}\notin\mathbb{Z}$, then

$$\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \longrightarrow \frac{g^m\,b^{\text{IntPart}[p]}\,(b\,x^n)^{\text{FracPart}[p]}}{x^{n\,\text{FracPart}[p]}}\int x^{m+n\,p}\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(b_.*x_^n_.)^p_*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  g^m*b^IntPart[p]*(b*x^n)^FracPart[p]/x^(n*FracPart[p])*Int[x^(m+n*p)*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{b,c,d,e,f,g,m,n,p,q,r},x] && (IntegerQ[m] || GtQ[g,0]) && Not[IntegerQ[Simplify[(m+1)/n]]]
```

2: $\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(g\,x)^m}{x^m} = 0$

Rule 1.1.3.6.0.2: If $m \notin \mathbb{Z}$, then

$$\int (g\,x)^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \longrightarrow \frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$$

Program code:

```
Int[(g_*x_)^m_*(b_.*x_^n_.)^p_*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  g^IntPart[m]*(g*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{b,c,d,e,f,g,m,n,p,q,r},x] && Not[IntegerQ[m]]
```

1: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $p+2\in\mathbb{Z}^+\land q\in\mathbb{Z}^+\land r\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.6.1: If $p+2\in\mathbb{Z}^+\land q\in\mathbb{Z}^+\land r\in\mathbb{Z}^+$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r,\,x\right]dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  Int[ExpandIntegrand[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n},x] && IGtQ[p,-2] && IGtQ[q,0] && IGtQ[r,0]
```

2: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $m-n+1==0$

Derivation: Integration by substitution

Basis: $x^{n-1}\,F[x^n] == \frac{1}{n}\,\text{Subst}[F[x],\,x,\,x^n]\,\partial_x x^n$

Rule 1.1.3.6.2: If $m-n+1==0$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int (a+b\,x)^p\,(c+d\,x)^q\,(e+f\,x)^r\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  1/n*Subst[Int[(a+b*x)^p*(c+d*x)^q*(e+f*x)^r,x],x,x^n] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && EqQ[m-n+1,0]
```

3: $\int x^m (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$ when $(p \mid q \mid r) \in \mathbb{Z} \land n<0$

Derivation: Algebraic expansion

Basis: If $(p \mid q \mid r) \in \mathbb{Z}$, then
$(a+b x^n)^p (c+d x^n)^q (e+f x^n)^r == x^{n (p+q+r)} (b+a x^{-n})^p (d+c x^{-n})^q (f+e x^{-n})^r$

Rule 1.1.3.6.3: If $(p \mid q \mid r) \in \mathbb{Z} \land n<0$, then

$$\int x^m (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx \longrightarrow \int x^{m+n (p+q+r)} (b+a x^{-n})^p (d+c x^{-n})^q (f+e x^{-n})^r \, dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  Int[x^(m+n*(p+q+r))*(b+a*x^(-n))^p*(d+c*x^(-n))^q*(f+e*x^(-n))^r,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IntegersQ[p,q,r] && NegQ[n]
```

4. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\partial_x x^n$

Note: If $n\in\mathbb{Z}\,\wedge\,\frac{m+1}{n}\in\mathbb{Z}$, then $m\in\mathbb{Z}$, and $(e\,x)^m$ automatically evaluates to $e^m\,x^m$.

Rule 1.1.3.6.4.1: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\;\to\;\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,(a+b\,x)^p\,(c+d\,x)^q\,(e+f\,x)^r\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*x)^p*(c+d*x)^q*(e+f*x)^r,x],x,x^n] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(g\,x)^m}{x^m} == 0$

Basis: $\frac{(g\,x)^m}{x^m} == \frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.6.4.2: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \longrightarrow \frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$$

Program code:

```
Int[(g_*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  g^IntPart[m]*(g*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r},x] && IntegerQ[Simplify[(m+1)/n]]
```

5. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}$

1. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}^+$

1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}^+\land m\in\mathbb{Z}\land \text{GCD}[m+1,n]\neq 1$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\ \land\ m\in\mathbb{Z}$, let $k=\text{GCD}[m+1,\ n]$, then $x^m\,F[x^n]=\frac{1}{k}\,\text{Subst}\left[x^{\frac{m+1}{k}-1}\,F\left[x^{n/k}\right],\,x,\,x^k\right]\,\partial_x x^k$

Rule 1.1.3.6.5.1.1: If $n\in\mathbb{Z}^+\land\ m\in\mathbb{Z}$, let $k=\text{GCD}[m+1,\ n]$, if $k\neq 1$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\;\longrightarrow\;\frac{1}{k}\,\text{Subst}\left[\int x^{\frac{m+1}{k}-1}\,\left(a+b\,x^{n/k}\right)^p\,\left(c+d\,x^{n/k}\right)^q\,\left(e+f\,x^{n/k}\right)^r\,dx,\,x,\,x^k\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  With[{k=GCD[m+1,n]},
  1/k*Subst[Int[x^((m+1)/k-1)*(a+b*x^(n/k))^p*(c+d*x^(n/k))^q*(e+f*x^(n/k))^r,x],x,x^k] /;
 k≠1] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && IGtQ[n,0] && IntegerQ[m]
```

2: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}^+\wedge m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $(g\,x)^m\,F[x] = \frac{k}{g}\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{g}\right],\,x,\,(g\,x)^{1/k}\right]\,\partial_x\,(g\,x)^{1/k}$

Rule 1.1.3.6.5.1.2: If $n\in\mathbb{Z}^+\wedge m\in\mathbb{F}$, let $k=\text{Denominator}[m]$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\;\to\;\frac{k}{g}\,\text{Subst}\left[\int x^{k\,(m+1)-1}\left(a+\frac{b\,x^{k\,n}}{g^n}\right)^p\left(c+\frac{d\,x^{k\,n}}{g^n}\right)^q\left(e+\frac{f\,x^{k\,n}}{g^n}\right)^r dx,\,x,\,(g\,x)^{1/k}\right]$$

Program code:

```
Int[(g_.*x_)^m_*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_)^r_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/g*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(k*n)/g^n)^p*(c+d*x^(k*n)/g^n)^q*(e+f*x^(k*n)/g^n)^r,x],x,(g*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,p,q,r},x] && IGtQ[n,0] && FractionQ[m]
```

3. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $n\in\mathbb{Z}^+$

1. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $n\in\mathbb{Z}^+\wedge p<-1$

1: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $n\in\mathbb{Z}^+\wedge p<-1\wedge q>0$

Derivation: Binomial product recurrence 1

Rule 1.1.3.6.5.1.3.1.1: If $n\in\mathbb{Z}^+\wedge p<-1\wedge q>0$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx\;\to$$
$$-\frac{(b\,e-a\,f)\,(g\,x)^{m+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q}{a\,b\,g\,n\,(p+1)}+\frac{1}{a\,b\,n\,(p+1)}\,\cdot$$

$$\int (g\,x)^m \left(a+b\,x^n\right)^{p+1} \left(c+d\,x^n\right)^{q-1} \left(c\left(b\,e\,n\,(p+1)+\left(b\,e-a\,f\right)(m+1)\right)+d\left(b\,e\,n\,(p+1)+\left(b\,e-a\,f\right)(m+n\,q+1)\right)x^n\right)dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  -(b*e-a*f)*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*b*g*n*(p+1)) +
  1/(a*b*n*(p+1))*Int[(g*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*
    Simp[c*(b*e*n*(p+1)+(b*e-a*f)*(m+1))+d*(b*e*n*(p+1)+(b*e-a*f)*(m+n*q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && IGtQ[n,0] && LtQ[p,-1] && GtQ[q,0] && Not[EqQ[q,1] && SimplerQ[b*c-a*d,b*e-a*f]]
```

2: $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right)dx$ when $n\in\mathbb{Z}^+\land p<-1\land m-n+1>0$

Derivation: Binomial product recurrence 3a

Rule 1.1.3.6.5.1.3.1.2: If $n\in\mathbb{Z}^+\land p<-1\land m-n+1>0$, then

$$\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right)dx \longrightarrow$$

$$\frac{g^{n-1}\left(b\,e-a\,f\right)(g\,x)^{m-n+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q+1}}{b\,n\left(b\,c-a\,d\right)(p+1)}-\frac{g^n}{b\,n\left(b\,c-a\,d\right)(p+1)}\cdot$$

$$\int (g\,x)^{m-n}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^q\left(c\left(b\,e-a\,f\right)(m-n+1)+\left(d\left(b\,e-a\,f\right)(m+n\,q+1)-b\,n\left(c\,f-d\,e\right)(p+1)\right)x^n\right)dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_),x_Symbol] :=
  g^(n-1)*(b*e-a*f)*(g*x)^(m-n+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(b*n*(b*c-a*d)*(p+1)) -
  g^n/(b*n*(b*c-a*d)*(p+1))*Int[(g*x)^(m-n)*(a+b*x^n)^(p+1)*(c+d*x^n)^q*
    Simp[c*(b*e-a*f)*(m-n+1)+(d*(b*e-a*f)*(m+n*q+1)-b*n*(c*f-d*e)*(p+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,q},x] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m-n+1,0]
```

3: $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right) dx$ when $n \in \mathbb{Z}^+ \land p < -1$

Derivation: Binomial product recurrence 3b

Rule 1.1.3.6.5.1.3.1.3: If $n \in \mathbb{Z}^+ \land p < -1$, then

$$\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right) dx \longrightarrow$$
$$-\frac{\left(b\,e-a\,f\right)\,(g\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q+1}}{a\,g\,n\left(b\,c-a\,d\right)(p+1)}+\frac{1}{a\,n\left(b\,c-a\,d\right)(p+1)}\cdot$$
$$\int (g\,x)^m \left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^q \left(c\left(b\,e-a\,f\right)(m+1)+e\,n\left(b\,c-a\,d\right)(p+1)+d\left(b\,e-a\,f\right)(m+n\,(p+q+2)+1)\,x^n\right) dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_),x_Symbol] :=
  -(b*e-a*f)*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*g*n*(b*c-a*d)*(p+1)) +
  1/(a*n*(b*c-a*d)*(p+1))*Int[(g*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^q*
    Simp[c*(b*e-a*f)*(m+1)+e*n*(b*c-a*d)*(p+1)+d*(b*e-a*f)*(m+n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,q},x] && IGtQ[n,0] && LtQ[p,-1]
```

2. $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right) dx$ when $n \in \mathbb{Z}^+ \land q > 0$

1: $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right) dx$ when $n \in \mathbb{Z}^+ \land q > 0 \land m < -1$

Derivation: Binomial product recurrence 2a

Rule 1.1.3.6.5.1.3.2.1: If $n \in \mathbb{Z}^+ \land q > 0 \land m < -1$, then

$$\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right) dx \longrightarrow$$
$$\frac{e\,(g\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^q}{a\,g\,(m+1)}-\frac{1}{a\,g^n\,(m+1)}\cdot$$

$$\int (g\,x)^{m+n}\,(a+b\,x^n)^p\,(c+d\,x^n)^{q-1}\,\left(c\,(b\,e-a\,f)\,(m+1)+e\,n\,(b\,c\,(p+1)+a\,d\,q)+d\,\left((b\,e-a\,f)\,(m+1)+b\,e\,n\,(p+q+1)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(g_.*x_)^m_*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  e*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*g*(m+1)) -
  1/(a*g^n*(m+1))*Int[(g*x)^(m+n)*(a+b*x^n)^p*(c+d*x^n)^(q-1)*
    Simp[c*(b*e-a*f)*(m+1)+e*n*(b*c*(p+1)+a*d*q)+d*((b*e-a*f)*(m+1)+b*e*n*(p+q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && IGtQ[n,0] && GtQ[q,0] && LtQ[m,-1] && Not[EqQ[q,1] && SimplerQ[e+f*x^n,c+d*x^n]]
```

2: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $n\in\mathbb{Z}^+\land q>0$

Derivation: Binomial product recurrence 2b

Rule 1.1.3.6.5.1.3.2.2: If $n\in\mathbb{Z}^+\land q>0$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx\longrightarrow$$

$$\frac{f\,(g\,x)^{m+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q}{b\,g\,(m+n\,(p+q+1)+1)}+\frac{1}{b\,(m+n\,(p+q+1)+1)}\,\cdot$$

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^{q-1}\,\left(c\,\left((b\,e-a\,f)\,(m+1)+b\,e\,n\,(p+q+1)\right)+\left(d\,(b\,e-a\,f)\,(m+1)+f\,n\,q\,(b\,c-a\,d)+b\,e\,d\,n\,(p+q+1)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  f*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(b*g*(m+n*(p+q+1)+1)) +
  1/(b*(m+n*(p+q+1)+1))*Int[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^(q-1)*
    Simp[c*((b*e-a*f)*(m+1)+b*e*n*(p+q+1))+(d*(b*e-a*f)*(m+1)+f*n*q*(b*c-a*d)+b*e*d*n*(p+q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && IGtQ[n,0] && GtQ[q,0] && Not[EqQ[q,1] && SimplerQ[e+f*x^n,c+d*x^n]]
```

3: $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right) dx$ when $n \in \mathbb{Z}^+ \land m > n-1$

Derivation: Binomial product recurrence 4a

Rule 1.1.3.6.5.1.3.3: If $n \in \mathbb{Z}^+ \land m > n-1$, then

$$\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right) dx \longrightarrow$$

$$\frac{f\,g^{n-1}\,(g\,x)^{m-n+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q+1}}{b\,d\,(m+n\,(p+q+1)+1)} - \frac{g^n}{b\,d\,(m+n\,(p+q+1)+1)}\,.$$

$$\int (g\,x)^{m-n}\left(a+b\,x^n\right)^p\left(c+d\,x^n\right)^q \left(a\,f\,c\,(m-n+1)+\left(a\,f\,d\,(m+n\,q+1)+b\left(f\,c\,(m+n\,p+1)-e\,d\,(m+n\,(p+q+1)+1)\right)\right)x^n\right) dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  f*g^(n-1)*(g*x)^(m-n+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(b*d*(m+n*(p+q+1)+1)) -
  g^n/(b*d*(m+n*(p+q+1)+1))*Int[(g*x)^(m-n)*(a+b*x^n)^p*(c+d*x^n)^q*
    Simp[a*f*c*(m-n+1)+(a*f*d*(m+n*q+1)+b*(f*c*(m+n*p+1)-e*d*(m+n*(p+q+1)+1)))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,p,q},x] && IGtQ[n,0] && GtQ[m,n-1]
```

4: $\int (g\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,\left(e+f\,x^n\right)\,dx$ when $n\in\mathbb{Z}^+\land m<-1$

Derivation: Binomial product recurrence 4b

Rule 1.1.3.6.5.1.3.4: If $n\in\mathbb{Z}^+\land m<-1$, then

$$\int (g\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,\left(e+f\,x^n\right)\,dx\;\longrightarrow$$

$$\frac{e\,(g\,x)^{m+1}\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^{q+1}}{a\,c\,g\,(m+1)}+\frac{1}{a\,c\,g^n\,(m+1)}\cdot$$

$$\int (g\,x)^{m+n}\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,\left(a\,f\,c\,(m+1)-e\,\left(b\,c+a\,d\right)\,(m+n+1)-e\,n\,\left(b\,c\,p+a\,d\,q\right)-b\,e\,d\,(m+n\,(p+q+2)+1)\,x^n\right)\,dx$$

Program code:

```
Int[(g_.*x_)^m_*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  e*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*c*g*(m+1)) +
  1/(a*c*g^n*(m+1))*Int[(g*x)^(m+n)*(a+b*x^n)^p*(c+d*x^n)^q*
    Simp[a*f*c*(m+1)-e*(b*c+a*d)*(m+n+1)-e*n*(b*c*p+a*d*q)-b*e*d*(m+n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,p,q},x] && IGtQ[n,0] && LtQ[m,-1]
```

5: $\int \frac{(g\,x)^m\,(a+b\,x^n)^p\,(e+f\,x^n)}{c+d\,x^n}\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.6.5.1.3.5: If $n\in\mathbb{Z}^+$, then

$$\int \frac{(g\,x)^m\,(a+b\,x^n)^p\,(e+f\,x^n)}{c+d\,x^n}\,dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{(g\,x)^m\,(a+b\,x^n)^p\,(e+f\,x^n)}{c+d\,x^n},\,x\right]dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(e_+f_.*x_^n_)/(c_+d_.*x_^n_),x_Symbol] :=
  Int[ExpandIntegrand[(g*x)^m*(a+b*x^n)^p*(e+f*x^n)/(c+d*x^n),x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && IGtQ[n,0]
```

6: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.6.5.1.3.6: If $n\in\mathbb{Z}^+$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx \longrightarrow$$
$$e\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx+\frac{f}{e^n}\int (g\,x)^{m+n}\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  e*Int[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^q,x] +
  f/e^n*Int[(g*x)^(m+n)*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p,q},x] && IGtQ[n,0]
```

4: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}^+\land r\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.6.5.1.4: If $n\in\mathbb{Z}^+\land\ r\in\mathbb{Z}^+$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\,\longrightarrow$$
$$e\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^{r-1}\,dx+\frac{f}{e^n}\int (g\,x)^{m+n}\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^{r-1}\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  e*Int[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^(r-1),x] +
  f/e^n*Int[(g*x)^(m+n)*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^(r-1),x] /;
FreeQ[{a,b,c,d,e,f,g,m,p,q},x] && IGtQ[n,0] && IGtQ[r,0]
```

2. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}^-$

   1. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}^-\land m\in\mathbb{Q}$

      1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{Z}^-\land m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $F[x] == -\text{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule 1.1.3.6.5.2.1.1: If $n\in\mathbb{Z}^-\land m\in\mathbb{Z}$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\;\longrightarrow\;-\text{Subst}\left[\int\frac{(a+b\,x^{-n})^p\,(c+d\,x^{-n})^q\,(e+f\,x^{-n})^r}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n))^p*(c+d*x^(-n))^q*(e+f*x^(-n))^r/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && ILtQ[n,0] && IntegerQ[m]
```

2: $\int (g x)^m (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$ when $n \in \mathbb{Z}^- \wedge m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z} \wedge k > 1$, then $(g x)^m F[x^n] = -\frac{k}{g} \text{Subst}\left[\frac{F[g^{-n} x^{-k n}]}{x^{k (m+1)+1}}, x, \frac{1}{(g x)^{1/k}}\right] \partial_x \frac{1}{(g x)^{1/k}}$

Rule 1.1.3.6.5.2.1.2: If $n \in \mathbb{Z}^- \wedge m \in \mathbb{F}$, let $k = \text{Denominator}[m]$, then

$$\int (g x)^m (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx \rightarrow -\frac{k}{g} \text{Subst}\left[\int \frac{(a+b g^{-n} x^{-k n})^p (c+d g^{-n} x^{-k n})^q (e+f g^{-n} x^{-k n})^r}{x^{k (m+1)+1}} \, dx, x, \frac{1}{(g x)^{1/k}}\right]$$

Program code:

```
Int[(g_.*x_)^m_*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/g*Subst[Int[(a+b*g^(-n)*x^(-k*n))^p*(c+d*g^(-n)*x^(-k*n))^q*(e+f*g^(-n)*x^(-k*n))^r/x^(k*(m+1)+1),x],x,1/(g*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,p,q,r},x] && ILtQ[n,0] && FractionQ[m]
```

2: $\int (g x)^m (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$ when $n \in \mathbb{Z}^- \wedge m \notin \mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \left((g x)^m (x^{-1})^m\right) = 0$

Basis: $F[x] = -\text{Subst}\left[\frac{F[x^{-1}]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule 1.1.3.6.5.2.2: If $n \in \mathbb{Z}^- \wedge m \notin \mathbb{Q}$, then

$$\int (g x)^m (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx \rightarrow (g x)^m (x^{-1})^m \int \frac{(a+b x^n)^p (c+d x^n)^q (e+f x^n)^r}{(x^{-1})^m} \, dx$$

$$\rightarrow -(g\,x)^m\,(x^{-1})^m\,\text{Subst}\left[\int\frac{(a+b\,x^{-n})^p\,(c+d\,x^{-n})^q\,(e+f\,x^{-n})^r}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(g_.*x_)^m_*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  -(g*x)^m*(x^(-1))^m*Subst[Int[(a+b*x^(-n))^p*(c+d*x^(-n))^q*(e+f*x^(-n))^r/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,f,g,m,p,q,r},x] && ILtQ[n,0] && Not[RationalQ[m]]
```

6. $\int(g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{F}$

1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $x^m\,F[x^n]=k\,\text{Subst}\left[x^{k\,(m+1)-1}\,F[x^{k\,n}],\,x,\,x^{1/k}\right]\partial_x x^{1/k}$

Rule 1.1.3.6.6.1: If $n\in\mathbb{F}$, let $k=\text{Denominator}[n]$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\rightarrow k\,\text{Subst}\left[\int x^{k\,(m+1)-1}\,(a+b\,x^{k\,n})^p\,(c+d\,x^{k\,n})^q\,(e+f\,x^{k\,n})^r\,dx,\,x,\,x^{1/k}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(k*n))^p*(c+d*x^(k*n))^q*(e+f*x^(k*n))^r,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,m,p,q,r},x] && FractionQ[n]
```

2: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n\in\mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{(g\,x)^m}{x^m}==0$

Basis: $\frac{(g\,x)^m}{x^m}==\frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.6.6.2: If $n\in\mathbb{F}$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\,\longrightarrow\,\frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$$

Program code:

```
Int[(g_*x_)^m_*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  g^IntPart[m]*(g*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p,q,r},x] && FractionQ[n]
```

7. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $\frac{n}{m+1}\in\mathbb{Z}$

1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1}\in\mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{m+1}\,\text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right],\,x,\,x^{m+1}\right]\,\partial_x\,x^{m+1}$

Rule 1.1.3.6.7.1: If $\frac{n}{m+1}\in\mathbb{Z}$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx\,\longrightarrow\,\frac{1}{m+1}\,\text{Subst}\left[\int\left(a+b\,x^{\frac{n}{m+1}}\right)^p\,\left(c+d\,x^{\frac{n}{m+1}}\right)^q\,\left(e+f\,x^{\frac{n}{m+1}}\right)^r\,dx,\,x,\,x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*x^Simplify[n/(m+1)])^p*(c+d*x^Simplify[n/(m+1)])^q*(e+f*x^Simplify[n/(m+1)])^r,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && IntegerQ[Simplify[n/(m+1)]]
```

2: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(g\,x)^m}{x^m} == 0$

Basis: $\frac{(g\,x)^m}{x^m} == \frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.6.7.2: If $\frac{n}{m+1}\in\mathbb{Z}$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \longrightarrow \frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$$

Program code:

```
Int[(g_*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  g^IntPart[m]*(g*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r},x] && IntegerQ[Simplify[n/(m+1)]]
```

8. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$

1. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $p<-1$

1: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $p<-1\wedge q>0$

Derivation: Binomial product recurrence 1

Rule 1.1.3.6.8.1.1: If $p<-1\wedge q>0$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx \longrightarrow$$
$$-\frac{(b\,e-a\,f)\,(g\,x)^{m+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q}{a\,b\,g\,n\,(p+1)}+\frac{1}{a\,b\,n\,(p+1)}\,\cdot$$

$$\int (g\,x)^m \left(a+b\,x^n\right)^{p+1} \left(c+d\,x^n\right)^{q-1} \left(c\,\left(b\,e\,n\,(p+1)+\left(b\,e-a\,f\right)\,(m+1)\right)+d\,\left(b\,e\,n\,(p+1)+\left(b\,e-a\,f\right)\,(m+n\,q+1)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  -(b*e-a*f)*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*b*g*n*(p+1)) +
  1/(a*b*n*(p+1))*Int[(g*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*
    Simp[c*(b*e*n*(p+1)+(b*e-a*f)*(m+1))+d*(b*e*n*(p+1)+(b*e-a*f)*(m+n*q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n},x] && LtQ[p,-1] && GtQ[q,0] && Not[EqQ[q,1] && SimplerQ[b*c-a*d,b*e-a*f]]
```

2: $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right)\,dx$ when $p<-1$

Derivation: Binomial product recurrence 3b

Rule 1.1.3.6.8.1.2: If $p<-1$, then

$$\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right)\,dx \longrightarrow$$

$$-\frac{\left(b\,e-a\,f\right)\,(g\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}\left(c+d\,x^n\right)^{q+1}}{a\,g\,n\,\left(b\,c-a\,d\right)\,(p+1)}+\frac{1}{a\,n\,\left(b\,c-a\,d\right)\,(p+1)}\cdot$$

$$\int (g\,x)^m \left(a+b\,x^n\right)^{p+1} \left(c+d\,x^n\right)^q \left(c\,\left(b\,e-a\,f\right)\,(m+1)+e\,n\,\left(b\,c-a\,d\right)\,(p+1)+d\,\left(b\,e-a\,f\right)\,(m+n\,(p+q+2)+1)\,x^n\right)\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_),x_Symbol] :=
  -(b*e-a*f)*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*g*n*(b*c-a*d)*(p+1)) +
  1/(a*n*(b*c-a*d)*(p+1))*Int[(g*x)^m*(a+b*x^n)^(p+1)*(c+d*x^n)^q*
    Simp[c*(b*e-a*f)*(m+1)+e*n*(b*c-a*d)*(p+1)+d*(b*e-a*f)*(m+n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && LtQ[p,-1]
```

2: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $q>0$

Derivation: Binomial product recurrence 2b

Rule 1.1.3.6.8.2: If $q>0$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx\;\longrightarrow$$

$$\frac{f\,(g\,x)^{m+1}\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q}{b\,g\,(m+n\,(p+q+1)+1)}+\frac{1}{b\,(m+n\,(p+q+1)+1)}\cdot$$

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^{q-1}\,\left(c\,\left((b\,e-a\,f)\,(m+1)+b\,e\,n\,(p+q+1)\right)+\left(d\,(b\,e-a\,f)\,(m+1)+f\,n\,q\,(b\,c-a\,d)+b\,e\,d\,n\,(p+q+1)\right)\,x^n\right)\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  f*(g*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(b*g*(m+n*(p+q+1)+1)) +
  1/(b*(m+n*(p+q+1)+1))*Int[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^(q-1)*
    Simp[c*((b*e-a*f)*(m+1)+b*e*n*(p+q+1))+(d*(b*e-a*f)*(m+1)+f*n*q*(b*c-a*d)+b*e*d*n*(p+q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && GtQ[q,0] && Not[EqQ[q,1]] && SimplerQ[e+f*x^n,c+d*x^n]
```

3: $\int \frac{(g\,x)^m\,(a+b\,x^n)^p\,(e+f\,x^n)}{c+d\,x^n}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Rule 1.1.3.6.8.3: If $b\,c-a\,d\neq 0$, then

$$\int \frac{(g\,x)^m\,(a+b\,x^n)^p\,(e+f\,x^n)}{c+d\,x^n}\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\Big[\frac{(g\,x)^m\,(a+b\,x^n)^p\,(e+f\,x^n)}{c+d\,x^n},\,x\Big]\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(e_+f_.*x_^n_)/(c_+d_.*x_^n_),x_Symbol] :=
  Int[ExpandIntegrand[(g*x)^m*(a+b*x^n)^p*(e+f*x^n)/(c+d*x^n),x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x]
```

4: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Rule 1.1.3.6.8.4: If $b\,c-a\,d\neq 0$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)\,dx\;\longrightarrow\;e\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx+\frac{f\,(g\,x)^m}{x^m}\int x^{m+n}\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_),x_Symbol] :=
  e*Int[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^q,x] +
  f*(g*x)^m/x^m*Int[x^(m+n)*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q},x]
```

9. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx$

1. $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx$

1: $\int x^m\,(a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx$ when $q \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $q \in \mathbb{Z}$, then $(c+d\,x^{-n})^q == x^{-n\,q}\,(d+c\,x^n)^q$

Rule 1.1.3.6.9.1.1: If $q \in \mathbb{Z}$, then

$$\int x^m\,(a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx \longrightarrow \int x^{m-n\,q}\,(a+b\,x^n)^p\,(d+c\,x^n)^q\,(e+f\,x^n)^r\,dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.*(e_+f_.*x_^n_.)^r_.,x_Symbol] :=
  Int[x^(m-n*q)*(a+b*x^n)^p*(d+c*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,r},x] && EqQ[mn,-n] && IntegerQ[q]
```

2: $\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \left(e+f\,x^n\right)^r dx$ when $p\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $p\in\mathbb{Z}$, then $(a+b\,x^n)^p == x^{n\,p}\,(b+a\,x^{-n})^p$

Rule 1.1.3.6.9.2: If $p\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \left(e+f\,x^n\right)^r dx \longrightarrow \int x^{m+n\,(p+r)} \left(b+a\,x^{-n}\right)^p \left(c+d\,x^{-n}\right)^q \left(f+e\,x^{-n}\right)^r dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.*(e_+f_.*x_^n_.)^r_.,x_Symbol] :=
  Int[x^(m+n*(p+r))*(b+a*x^(-n))^p*(c+d*x^(-n))^q*(f+e*x^(-n))^r,x] /;
FreeQ[{a,b,c,d,e,f,m,n,q},x] && EqQ[mn,-n] && IntegerQ[p] && IntegerQ[r]
```

3: $\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \left(e+f\,x^n\right)^r dx$ when $q \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,q}\,\left(c+d\,x^{-n}\right)^q}{\left(d+c\,x^n\right)^q} == 0$

Basis: $\frac{x^{n\,q}\,\left(c+d\,x^{-n}\right)^q}{\left(d+c\,x^n\right)^q} == \frac{x^{n\,\text{FracPart}[q]}\,\left(c+d\,x^{-n}\right)^{\text{FracPart}[q]}}{\left(d+c\,x^n\right)^{\text{FracPart}[q]}}$

Rule 1.1.3.6.9.3: If $q \notin \mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \left(e+f\,x^n\right)^r dx \to \frac{x^{n\,\text{FracPart}[q]}\,\left(c+d\,x^{-n}\right)^{\text{FracPart}[q]}}{\left(d+c\,x^n\right)^{\text{FracPart}[q]}} \int x^{m-n\,q} \left(a+b\,x^n\right)^p \left(d+c\,x^n\right)^q \left(e+f\,x^n\right)^r dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_*(e_+f_.*x_^n_.)^r_.,x_Symbol] :=
  x^(n*FracPart[q])*(c+d*x^(-n))^FracPart[q]/(d+c*x^n)^FracPart[q]*Int[x^(m-n*q)*(a+b*x^n)^p*(d+c*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && EqQ[mn,-n] && Not[IntegerQ[q]]
```

2: $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \left(e+f\,x^n\right)^r dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(g\,x)^m}{x^m} == 0$

Basis: $\frac{(g\,x)^m}{x^m} == \frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule 1.1.3.6.9.2:

$$\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \left(e+f\,x^n\right)^r dx \longrightarrow \frac{g^{\text{IntPart}[m]}\,(g\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a+b\,x^n\right)^p \left(c+d\,x^{-n}\right)^q \left(e+f\,x^n\right)^r dx$$

Program code:

```
Int[(g_*x_)^m_*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.*(e_+f_.*x_^n_.)^r_.,x_Symbol] :=
  g^IntPart[m]*(g*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p*(c+d*x^(-n))^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r},x] && EqQ[mn,-n]
```

X: $\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right)^r dx$

Rule 1.1.3.6.X:

$$\int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right)^r dx \longrightarrow \int (g\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right)^q \left(e+f\,x^n\right)^r dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  Unintegrable[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r},x]
```

S: $\int u^m (a+b v^n)^p (c+d v^n)^q (e+f v^n)^r \, dx$ when $v == h + i x \land u == g v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u == g v$, then $\partial_x \frac{u^m}{v^m} == 0$

Rule 1.1.3.6.S: If $v == h + i x \land u == g v$, then

$$\int u^m (a+b v^n)^p (c+d v^n)^q (e+f v^n)^r \, dx \longrightarrow \frac{u^m}{i v^m} \text{Subst}\left[\int x^m (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx, x, v\right]$$

Program code:

```
Int[u_^m_.*(a_.+b_.*v_^n_)^p_.*(c_.+d_.*v_^n_)^q_.*(e_+f_.*v_^n_)^r_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x],x,v] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && LinearPairQ[u,v,x]
```

Rules for integrands of the form $(g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r$

1. $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx$ when $e_2\,f_1+e_1\,f_2==0$

1: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx$ when $e_2\,f_1+e_1\,f_2==0\,\land\,(r\in\mathbb{Z}\,\lor\,e_1>0\,\land\,e_2>0)$

Derivation: Algebraic simplification

Basis: If $e_2\,f_1+e_1\,f_2==0\,\land\,(r\in\mathbb{Z}\,\lor\,e_1>0\,\land\,e_2>0)$, then $\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r==\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r$

Rule: If $e_2\,f_1+e_1\,f_2==0\,\land\,(r\in\mathbb{Z}\,\lor\,e_1>0\,\land\,e_2>0)$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx\,\longrightarrow\,\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e1_+f1_.*x_^n2_.)^r_.*(e2_+f2_.*x_^n2_.)^r_.,x_Symbol] :=
  Int[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^q*(e1*e2+f1*f2*x^n)^r,x] /;
FreeQ[{a,b,c,d,e1,f1,e2,f2,g,m,n,p,q,r},x] && EqQ[n2,n/2] && EqQ[e2*f1+e1*f2,0] && (IntegerQ[r] || GtQ[e1,0] && GtQ[e2,0])
```

2: $\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx$ when $e_2\,f_1+e_1\,f_2==0$

Derivation: Piecewise constant extraction

Basis: If $e_2\,f_1+e_1\,f_2==0$, then $\partial_x\,\frac{\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r}{\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r}==0$

Rule: If $e_2\,f_1+e_1\,f_2==0$, then

$$\int (g\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx\,\longrightarrow$$

$$\frac{\left(e_1+f_1\,x^{n/2}\right)^{\text{FracPart}[r]}\left(e_2+f_2\,x^{n/2}\right)^{\text{FracPart}[r]}}{\left(e_1\,e_2+f_1\,f_2\,x^n\right)^{\text{FracPart}[r]}}\int(g\,x)^m\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^q\,\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e1_+f1_.*x_^n2_.)^r_.*(e2_+f2_.*x_^n2_.)^r_.,x_Symbol] :=
  (e1+f1*x^(n/2))^FracPart[r]*(e2+f2*x^(n/2))^FracPart[r]/(e1*e2+f1*f2*x^n)^FracPart[r]*
    Int[(g*x)^m*(a+b*x^n)^p*(c+d*x^n)^q*(e1*e2+f1*f2*x^n)^r,x] /;
FreeQ[{a,b,c,d,e1,f1,e2,f2,g,m,n,p,q,r},x] && EqQ[n2,n/2] && EqQ[e2*f1+e1*f2,0]
```