

Rules for integrands of the form $(dx)^m P_q[x] (a + bx^n + cx^{2n})^p$

1: $\int x^m P_q[x^n] (a + bx^n + cx^{2n})^p dx$ when $m - n + 1 = 0$

Derivation: Integration by substitution

Basis: $x^{n-1} F[x^n] = \frac{1}{n} \text{Subst}[F[x], x, x^n] \partial_x x^n$

Rule: If $m - n + 1 = 0$, then

$$\int x^m P_q[x^n] (a + bx^n + cx^{2n})^p dx \rightarrow \frac{1}{n} \text{Subst} \left[\int P_q[x] (a + bx + cx^2)^p dx, x, x^n \right]$$

Program code:

```
Int[x^m_*Pq_*(a+b_*x_^n+c_*x_^n2_)^p_,x_Symbol] :=
  1/n*Subst[Int[SubstFor[x^n,Pq,x]*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && EqQ[Simplify[m-n+1],0]
```

2: $\int (dx)^m P_q[x] (a + bx^n + cx^{2n})^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (dx)^m P_q[x] (a + bx^n + cx^{2n})^p dx \rightarrow \int \text{ExpandIntegrand}[(dx)^m P_q[x] (a + bx^n + cx^{2n})^p, x] dx$$

Program code:

```
Int[(d_*x_)^m_*Pq_*(a+b_*x_^n+c_*x_^n2_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*Pq*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && IGtQ[p,0]
```

3: $\int (gx)^m (d+ex^n+fx^{2n}) (a+bx^n+cx^{2n})^p dx$ when $a e (m+1) - b d (m+n(p+1)+1) = 0 \wedge a f (m+1) - c d (m+2n(p+1)+1) = 0 \wedge m \neq -1$

Rule: If $a e (m+1) - b d (m+n(p+1)+1) = 0 \wedge a f (m+1) - c d (m+2n(p+1)+1) = 0 \wedge m \neq -1$, then

$$\int (gx)^m (d+ex^n+fx^{2n}) (a+bx^n+cx^{2n})^p dx \rightarrow \frac{d (gx)^{m+1} (a+bx^n+cx^{2n})^{p+1}}{a g (m+1)}$$

Program code:

```
Int[(g_.**x_)^m_.*(d_+e_.**x_^n_.+f_.**x_^n2_.)*(a_+b_.**x_^n_.+c_.**x_^n2_.)^p_,x_Symbol] :=
  d*(g*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*g*(m+1)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[n2,2*n] && EqQ[a*e*(m+1)-b*d*(m+n*(p+1)+1),0] && EqQ[a*f*(m+1)-c*d*(m+2*n*(p+1)+1),0] && N
```

```
Int[(g_.**x_)^m_.*(d_+f_.**x_^n2_.)*(a_+b_.**x_^n_.+c_.**x_^n2_.)^p_,x_Symbol] :=
  d*(g*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*g*(m+1)) /;
FreeQ[{a,b,c,d,f,g,m,n,p},x] && EqQ[n2,2*n] && EqQ[m+n*(p+1)+1,0] && EqQ[c*d+a*f,0] && NeQ[m,-1]
```

4: $\int (dx)^m P_q(x) (a+bx^n+cx^{2n})^p dx$ when $b^2 - 4ac == 0 \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4ac == 0$, then $\partial_x \frac{(a+bx^n+cx^{2n})^p}{(b+2cx^n)^{2p}} == 0$

Basis: If $b^2 - 4ac == 0$, then $\frac{(a+bx^n+cx^{2n})^p}{(b+2cx^n)^{2p}} == \frac{(a+bx^n+cx^{2n})^{\text{FracPart}[p]}}{(4c)^{\text{IntPart}[p]} (b+2cx^n)^{2\text{FracPart}[p]}}$

Rule: If $b^2 - 4ac == 0 \wedge p \notin \mathbb{Z}$, then

$$\int (dx)^m P_q(x) (a+bx^n+cx^{2n})^p dx \rightarrow \frac{(a+bx^n+cx^{2n})^{\text{FracPart}[p]}}{(4c)^{\text{IntPart}[p]} (b+2cx^n)^{2\text{FracPart}[p]}} \int (dx)^m P_q(x) (b+2cx^n)^{2p} dx$$

Program code:

```
Int[(d.*x_)^m_.*Pq_*(a+b.*x_^n_.+c.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x^n)^(2*FracPart[p]))*Int[(d*x)^m*Pq*(b+2*c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```

$$5. \int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge \frac{m+1}{n} \in \mathbb{Z}$$

$$1: \int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge \frac{m+1}{n} \in \mathbb{Z}$$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n} \in \mathbb{Z}$, then $x^m F[x^n] = \frac{1}{n} \text{Subst}[x^{\frac{m+1}{n}-1} F[x], x, x^n] \partial_x x^n$

Note: If $n \in \mathbb{Z} \wedge \frac{m+1}{n} \in \mathbb{Z}$, then $m \in \mathbb{Z}$, and $(dx)^m$ automatically evaluates to $d^m x^m$.

Rule: If $b^2 - 4ac \neq 0 \wedge \frac{m+1}{n} \in \mathbb{Z}$, then

$$\int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \rightarrow \frac{1}{n} \text{Subst}\left[\int x^{\frac{m+1}{n}-1} P_q[x] (a+bx+cx^2)^p dx, x, x^n\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*SubstFor[x^n,Pq,x]*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx$ when $b^2 - 4ac \neq 0 \wedge \frac{m+1}{n} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(dx)^m}{x^m} = 0$

Rule: If $b^2 - 4ac \neq 0 \wedge \frac{m+1}{n} \in \mathbb{Z}$, then

$$\int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \rightarrow \frac{(dx)^m}{x^m} \int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d*x_)^m_.*Pq*(a+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[(m+1)/n]]
```

6: $\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx$ when $\text{PolynomialRemainder}[P_q[x], x, x] = 0$

Derivation: Algebraic simplification

Rule: If $\text{PolynomialRemainder}[P_q[x], x, x] = 0$, then

$$\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \rightarrow \frac{1}{d} \int (dx)^{m+1} \text{PolynomialQuotient}[P_q[x], x, x] (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq*(a+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/d*Int[(d*x)^(m+1)*PolynomialQuotient[Pq,x,x]*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,x,x],0]
```

$$7. \int \frac{(dx)^m (e + f x^{n/2} + g x^{3n/2} + h x^{2n})}{(a + b x^n + c x^{2n})^{3/2}} dx \text{ when } b^2 - 4ac = 0 \wedge 2m - n + 2 = 0 \wedge ce + ah = 0$$

$$1: \int \frac{x^m (e + f x^{n/2} + g x^{3n/2} + h x^{2n})}{(a + b x^n + c x^{2n})^{3/2}} dx \text{ when } b^2 - 4ac = 0 \wedge 2m - n + 2 = 0 \wedge ce + ah = 0$$

Rule: If $b^2 - 4ac = 0 \wedge 2m - n + 2 = 0 \wedge ce + ah = 0$, then

$$\int \frac{x^m (e + f x^{n/2} + g x^{3n/2} + h x^{2n})}{(a + b x^n + c x^{2n})^{3/2}} dx \rightarrow -\frac{2c(bf - 2ag) + 2h(b^2 - 4ac)x^{n/2} + 2c(2cf - bg)x^n}{cn(b^2 - 4ac)\sqrt{a + bx^n + cx^{2n}}}$$

Program code:

```
Int[x^m_.*(e+f_.*x^q_.+g_.*x^r_.+h_.*x^s_.)/(a+b_.*x^n_.+c_.*x^n2_.)^(3/2),x_Symbol] :=
  -(2*c*(b*f-2*a*g)+2*h*(b^2-4*a*c)*x^(n/2)+2*c*(2*c*f-b*g)*x^n)/(c*n*(b^2-4*a*c)*Sqrt[a+b*x^n+c*x^(2*n)]) /;
FreeQ[{a,b,c,e,f,g,h,m,n},x] && EqQ[n2,2*n] && EqQ[q,n/2] && EqQ[r,3*n/2] && EqQ[s,2*n] &&
  NeQ[b^2-4*a*c,0] && EqQ[2*m-n+2,0] && EqQ[c*e+a*h,0]
```

$$2: \int \frac{(dx)^m (e + f x^{n/2} + g x^{3n/2} + h x^{2n})}{(a + b x^n + c x^{2n})^{3/2}} dx \text{ when } b^2 - 4ac = 0 \wedge 2m - n + 2 = 0 \wedge ce + ah = 0$$

Rule: If $b^2 - 4ac = 0 \wedge 2m - n + 2 = 0 \wedge ce + ah = 0$, then

$$\int \frac{(dx)^m (e + f x^{n/2} + g x^{3n/2} + h x^{2n})}{(a + b x^n + c x^{2n})^{3/2}} dx \rightarrow \frac{(dx)^m}{x^m} \int \frac{x^m (e + f x^{n/2} + g x^{3n/2} + h x^{2n})}{(a + b x^n + c x^{2n})^{3/2}} dx$$

Program code:

```
Int[(d*x_)^m.*(e_+f_.*x_^q_.+g_.*x_^r_.+h_.*x_^s_.)/(a_+b_.*x_^n_.+c_.*x_^n2_.)^(3/2),x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*(e+f*x^(n/2)+g*x^((3*n)/2)+h*x^(2*n))/(a+b*x^n+c*x^(2*n))^(3/2),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[n2,2*n] && EqQ[q,n/2] && EqQ[r,3*n/2] && EqQ[s,2*n] &&
NeQ[b^2-4*a*c,0] && EqQ[2*m-n+2,0] && EqQ[c*e+a*h,0]
```

$$8. \int (dx)^m P_q[x] (a + b x^n + c x^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}$$

$$1. \int (dx)^m P_q[x] (a + b x^n + c x^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+$$

$$1: \int x^m P_q[x] (a + b x^n + c x^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge p < -1 \wedge m \in \mathbb{Z}^-$$

Derivation: Algebraic expansion and trinomial recurrence 2b applied $n-1$ times

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge p < -1 \wedge m \in \mathbb{Z}^-$, let $Q_{q-2n}[x] = \text{PolynomialQuotient}[x^m P_q[x], a + b x^n + c x^{2n}, x]$ and $R_{2n-1}[x] = \text{PolynomialRemainder}[x^m P_q[x], a + b x^n + c x^{2n}, x]$, then

$$\int x^m P_q[x] (a + b x^n + c x^{2n})^p dx \rightarrow$$

$$\int R_{2n-1}[x] (a + b x^n + c x^{2n})^p dx + \int Q_{q-2n}[x] (a + b x^n + c x^{2n})^{p+1} dx \rightarrow$$

$$\begin{aligned}
& - \left(\left(x (a + b x^n + c x^{2n})^{p+1} \sum_{i=0}^{n-1} \left((b^2 - 2ac) R_{2n-1}[x, i] - ab R_{2n-1}[x, n+i] \right) x^i + c (b R_{2n-1}[x, i] - 2a R_{2n-1}[x, n+i]) x^{n+i} \right) \right) / \\
& \quad (a n (p+1) (b^2 - 4ac)) + \\
& \quad \frac{1}{a n (p+1) (b^2 - 4ac)} \int x^m (a + b x^n + c x^{2n})^{p+1} \left(a n (p+1) (b^2 - 4ac) x^{-m} Q_{q-2n}[x] + \right. \\
& \quad \sum_{i=0}^{n-1} \left((b^2 (n(p+1) + i + 1) - 2ac (2n(p+1) + i + 1)) R_{2n-1}[x, i] - ab (i+1) R_{2n-1}[x, n+i] \right) x^{i-m} + \\
& \quad \left. c (n(2p+3) + i + 1) (b R_{2n-1}[x, i] - 2a R_{2n-1}[x, n+i]) x^{n+i-m} \right) dx
\end{aligned}$$

Program code:

```

Int[x_^m_*Pq_*(a_+b_*x_^n_+c_*x_^n2_)^p_,x_Symbol] :=
Module[{q=Expon[Pq,x]},
Module[{Q=PolynomialQuotient[a*(b*c)^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n+c*x^(2*n),x],
R=PolynomialRemainder[a*(b*c)^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n+c*x^(2*n),x],i},
-x*(a+b*x^n+c*x^(2*n))^(p+1)/(a^2*n*(p+1)*(b^2-4*a*c)*(b*c)^(Floor[(q-1)/n]+1))*
Sum[(b^2-2*a*c)*Coeff[R,x,i]-a*b*Coeff[R,x,n+i])*x^i+
c*(b*Coeff[R,x,i]-2*a*Coeff[R,x,n+i])*x^(n+i),{i,0,n-1}] +
1/(a*n*(p+1)*(b^2-4*a*c)*(b*c)^(Floor[(q-1)/n]+1))*Int[x^m*(a+b*x^n+c*x^(2*n))^(p+1)*
ExpandToSum[n*(p+1)*(b^2-4*a*c)*x^(-m)*Q+
Sum[(b^2*(n*(p+1)+i+1)/a-2*c*(2*n*(p+1)+i+1))*Coeff[R,x,i]-b*(i+1)*Coeff[R,x,n+i])*x^(i-m)+
c*(n*(2*p+3)+i+1)*(b/a*Coeff[R,x,i]-2*Coeff[R,x,n+i])*x^(n+i-m),{i,0,n-1}],x,x]] /;
GeQ[q,2*n]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] && ILtQ[m,0]

```


$$2. \int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+$$

$$1: \int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z} \wedge \text{GCD}[m+1, n] \neq 1$$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z} \wedge m \in \mathbb{Z}$, let $g = \text{GCD}[m+1, n]$, then $x^m F[x^n] = \frac{1}{g} \text{Subst}[x^{\frac{m+1}{g}-1} F[x^{\frac{n}{g}}], x, x^g] \partial_x x^g$

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$, let $g = \text{GCD}[m+1, n]$, if $g \neq 1$, then

$$\int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \rightarrow \frac{1}{g} \text{Subst}\left[\int x^{\frac{m+1}{g}-1} P_q\left[x^{\frac{n}{g}}\right] (a+bx^{\frac{n}{g}}+cx^{\frac{2n}{g}})^p dx, x, x^g\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{g=GCD[m+1,n]},
    1/g*Subst[Int[x^((m+1)/g-1)*ReplaceAll[Pq,x->x^(1/g)]*(a+b*x^(n/g)+c*x^(2*n/g))^p,x],x,x^g] /;
    NeQ[g,1] /;
    FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IntegerQ[m]
```

2: $\int \frac{(dx)^m P_q[x^n]}{a+bx^n+cx^{2n}} dx$ when $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge \text{NiceSqrtQ}[b^2 - 4ac]$

Derivation: Algebraic expansion

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge \text{NiceSqrtQ}[b^2 - 4ac]$, then

$$\int \frac{(dx)^m P_q[x^n]}{a+bx^n+cx^{2n}} dx \rightarrow \int \text{ExpandIntegrand}\left[\frac{(dx)^m P_q[x^n]}{a+bx^n+cx^{2n}}, x\right] dx$$

Program code:

```
Int[(d.*x_)^m_.*Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_),x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*Pq/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && NiceSqrtQ[b^2-4*a*c]
```

3: $\int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx$ when $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge q \geq 2n \wedge m+q+2np+1 \neq 0$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A = 0$, $B = 1$ and $m = m - n$

Reference: G&R 2.104

Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge q \geq 2n \wedge m+q+2np+1 \neq 0$, then

$$\int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \rightarrow$$

$$\int (d x)^m \left(P_q[x^n] - P_q[x, q] x^q \right) (a+b x^n+c x^{2 n})^p dx + \frac{P_q[x, q]}{d^q} \int (d x)^{m+q} (a+b x^n+c x^{2 n})^p dx \rightarrow$$

$$\frac{P_q[x, q] (d x)^{m+q-2 n+1} (a+b x^n+c x^{2 n})^{p+1}}{c d^{q-2 n+1} (m+q+2 n p+1)} +$$

$$\int (d x)^m \left(P_q[x^n] - P_q[x, q] x^q - \frac{P_q[x, q] (a(m+q-2 n+1) x^{q-2 n}+b(m+q+n(p-1)+1) x^{q-n})}{c(m+q+2 n p+1)} \right) (a+b x^n+c x^{2 n})^p dx$$

Program code:

```
Int[(d.*x_)^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
      Pqq*(d*x)^(m+q-2*n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*d^(q-2*n+1)*(m+q+2*n*p+1)) +
      Int[(d*x)^m*ExpandToSum[Pq-Pqq*x^q-Pqq*(a*(m+q-2*n+1)*x^(q-2*n)+b*(m+q+n*(p-1)+1)*x^(q-n)]/(c*(m+q+2*n*p+1)),x]*
      (a+b*x^n+c*x^(2*n))^p,x]] /;
    GeQ[q,2*n] && NeQ[m+q+2*n*p+1,0] && (IntegerQ[2*p] || EqQ[n,1] && IntegerQ[4*p] || IntegerQ[p+(q+1)/(2*n)])] /;
    FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

$$\mathbf{3:} \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge \neg \text{PolynomialQ}[P_q[x], x^n]$$

Derivation: Algebraic expansion

Basis: If $n \in \mathbb{Z}^+$, then $P_q[x] = \sum_{j=0}^{q-1} x^j \sum_{k=0}^{(q-j)/n+1} P_q[x, j+kn] x^{kn}$

Note: This rule transform integrand into a sum of terms of the form $(dx)^k Q_r[x^n] (a+bx^n+cx^{2n})^p$.

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+ \wedge \neg \text{PolynomialQ}[P_q[x], x^n]$, then

$$\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \rightarrow \int \sum_{j=0}^{q-1} \frac{1}{d^j} (dx)^{m+j} \left(\sum_{k=0}^{(q-j)/n+1} P_q[x, j+kn] x^{kn} \right) (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d_.**x_)^m_.**Pq_*(a_+b_.**x_^n_+c_.**x_^n2_)^p_,x_Symbol] :=
Module[{q=Expon[Pq,x],j,k},
Int[Sum[1/d^j*(d*x)^(m+j)*Sum[Coeff[Pq,x,j+k*n]**x^(k*n),{k,0,(q-j)/n+1}]*
(a+b*x^n+c*x^(2*n))^p,{j,0,n-1}],x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[PolyQ[Pq,x^n]]
```

4: $\int \frac{(dx)^m P_q[x]}{a+bx^n+cx^{2n}} dx$ when $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^+$, then

$$\int \frac{(dx)^m P_q[x]}{a+bx^n+cx^{2n}} dx \rightarrow \int \text{RationalFunctionExpand}\left[\frac{(dx)^m P_q[x]}{a+bx^n+cx^{2n}}, x\right] dx$$

Program code:

```
Int[(d.*x_)^m_.*Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_.),x_Symbol] :=
  Int[RationalFunctionExpand[(d*x)^m*Pq/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

$$2. \int (dx)^m P_q(x) (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^-$$

$$1. \int (dx)^m P_q(x) (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^- \wedge m \in \mathbb{Q}$$

$$\text{1: } \int x^m P_q(x) (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^- \wedge m \in \mathbb{Z}$$

Derivation: Integration by substitution

$$\text{Basis: } F[x] = -\text{Subst} \left[\frac{F[x^{-1}]}{x^2}, x, \frac{1}{x} \right] \partial_x \frac{1}{x}$$

Note: $x^q P_q[x^{-1}]$ is a polynomial in x .

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^- \wedge m \in \mathbb{Z}$, then

$$\int x^m P_q(x) (a+bx^n+cx^{2n})^p dx \rightarrow -\text{Subst} \left[\int \frac{x^q P_q[x^{-1}] (a+bx^{-n}+cx^{-2n})^p}{x^{m+q+2}} dx, x, \frac{1}{x} \right]$$

Program code:

```
Int[x^m_.*Pq_*(a+_.*x_^n+_.*x_^2n_.)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    -Subst[Int[ExpandToSum[x^q*ReplaceAll[Pq,x->x^(-1)],x]*(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+q+2),x],x,1/x] /;
    FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && IntegerQ[m]
```

$$2: \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^- \wedge m \in \mathbb{F}$$

Derivation: Integration by substitution

Basis: If $g > 1$, then $(dx)^m F[x] = -\frac{g}{d} \text{Subst}\left[\frac{F[d^{-1}x^{-g}]}{x^{g(m+1)+1}}, x, \frac{1}{(dx)^{1/g}}\right] \partial_x \frac{1}{(dx)^{1/g}}$

Note: $x^{gq} P_q[d^{-1}x^{-g}]$ is a polynomial in x .

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^- \wedge m \in \mathbb{F}$, let $g = \text{Denominator}[m]$, then

$$\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \rightarrow -\frac{g}{d} \text{Subst}\left[\int \frac{x^{gq} P_q[d^{-1}x^{-g}] (a+bd^{-n}x^{-gn}+cd^{-2n}x^{-2gn})^p}{x^{g(m+q+1)+1}} dx, x, \frac{1}{(dx)^{1/g}}\right]$$

Program code:

```
Int[(d_.**x_)^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{g=Denominator[m],q=Expon[Pq,x]},
    -g/d*Subst[Int[ExpandToSum[x^(g*q)*ReplaceAll[Pq,x->d^(-1)*x^(-g)],x]*
      (a+b*d^(-n)*x^(-g*n)+c*d^(-2*n)*x^(-2*g*n))^p/x^(g*(m+q+1)+1),x],x,1/(d*x)^(1/g)]] /;
  FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && FractionQ[m]
```

$$\mathbf{2:} \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^- \wedge m \notin \mathbb{Q}$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: } \partial_x \left((dx)^m (x^{-1})^m \right) = 0$$

$$\text{Basis: } F[x] = -\text{Subst} \left[\frac{F[x^{-1}]}{x^2}, x, \frac{1}{x} \right] \partial_x \frac{1}{x}$$

Note: $x^q P_q[x^{-1}]$ is a polynomial in x .

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{Z}^- \wedge m \notin \mathbb{Q}$, then

$$\begin{aligned} \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx &\rightarrow (dx)^m (x^{-1})^m \int \frac{P_q[x] (a+bx^n+cx^{2n})^p}{(x^{-1})^m} dx \\ &\rightarrow - (dx)^m (x^{-1})^m \text{Subst} \left[\int \frac{x^q P_q[x^{-1}] (a+bx^{-n}+cx^{-2n})^p}{x^{m+q+2}} dx, x, \frac{1}{x} \right] \end{aligned}$$

Program code:

```
Int[(d.*x_)^m.*Pq.*(a+b.*x_^n+c.*x_^2n)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    -(d*x)^m*(x^(-1))^m*Subst[Int[ExpandToSum[x^q*ReplaceAll[Pq,x->x^(-1)],x]*(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+q+2),x],x,1/x]] /;
  FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && !LtQ[n,0] && !RationalQ[m]
```


$$9. \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{F}$$

$$1: \int x^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{F}$$

Derivation: Integration by substitution

Basis: If $g \in \mathbb{Z}^+$, then $x^m P_q[x] F[x^n] = g \text{Subst}[x^{g(m+1)-1} P_q[x^g] F[x^{g n}], x, x^{1/g}] \partial_x x^{1/g}$

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{F}$, let $g = \text{Denominator}[n]$, then

$$\int x^m P_q[x] (a+bx^n+cx^{2n})^p dx \rightarrow g \text{Subst}\left[\int x^{g(m+1)-1} P_q[x^g] (a+bx^{g n}+cx^{2g n})^p dx, x, x^{1/g}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[n]},
    g*Subst[Int[x^(g*(m+1)-1)*ReplaceAll[Pq,x->x^g]*(a+b*x^(g*n)+c*x^(2*g*n))^p,x],x,x^(1/g)]] /;
  FreeQ[{a,b,c,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

$$2. \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{F}$$

$$1. \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{F} \wedge m - \frac{1}{2} \in \mathbb{Z}$$

$$1: \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{F} \wedge m + \frac{1}{2} \in \mathbb{Z}^+$$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{dx}}{\sqrt{x}} = 0$

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{F} \wedge m + \frac{1}{2} \in \mathbb{Z}^+$, then

$$\int (dx)^m P_q(x) (a+bx^n+cx^{2n})^p dx \rightarrow \frac{d^{m-\frac{1}{2}} \sqrt{dx}}{\sqrt{x}} \int x^m P_q(x) (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d*x_)^m*Pq*(a+b_.**x_^n+c_.**x_^n2_.)^p_,x_Symbol] :=
  d^(m-1/2)*Sqrt[d*x]/Sqrt[x]*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n] && IGtQ[m+1/2,0]
```

2: $\int (dx)^m P_q(x) (a+bx^n+cx^{2n})^p dx$ when $b^2 - 4ac \neq 0 \wedge n \in \mathbb{F} \wedge m - \frac{1}{2} \in \mathbb{Z}^-$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{x}}{\sqrt{dx}} = 0$

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{F} \wedge m - \frac{1}{2} \in \mathbb{Z}^-$, then

$$\int (dx)^m P_q(x) (a+bx^n+cx^{2n})^p dx \rightarrow \frac{d^{m+\frac{1}{2}} \sqrt{x}}{\sqrt{dx}} \int x^m P_q(x) (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d*x_)^m*Pq*(a+b_.**x_^n+c_.**x_^n2_.)^p_,x_Symbol] :=
  d^(m+1/2)*Sqrt[x]/Sqrt[d*x]*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n] && ILtQ[m-1/2,0]
```

$$2: \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge n \in \mathbb{F}$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(dx)^m}{x^m} = 0$$

Rule: If $b^2 - 4ac \neq 0 \wedge n \in \mathbb{F}$, then

$$\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \rightarrow \frac{(dx)^m}{x^m} \int x^m P_q[x] (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d*x_)^m_*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

$$10. \int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge \frac{n}{m+1} \in \mathbb{Z}$$

$$1: \int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \text{ when } b^2 - 4ac \neq 0 \wedge \frac{n}{m+1} \in \mathbb{Z}$$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1} \in \mathbb{Z}$, then $x^m F[x^n] = \frac{1}{m+1} \text{Subst}[F[x^{\frac{n}{m+1}}], x, x^{m+1}] \partial_x x^{m+1}$

Rule: If $b^2 - 4ac \neq 0 \wedge \frac{n}{m+1} \in \mathbb{Z}$

$$\int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \rightarrow \frac{1}{m+1} \text{Subst}\left[\int P_q\left[x^{\frac{n}{m+1}}\right] (a+bx^{\frac{n}{m+1}}+cx^{\frac{2n}{m+1}})^p dx, x, x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^2n_.)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[ReplaceAll[SubstFor[x^n,Pq,x],x->x^Simplify[n/(m+1)]]*(a+b*x^Simplify[n/(m+1)]+c*x^Simplify[2*n/(m+1)])^p,x],
  FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx$ when $b^2 - 4ac \neq 0 \wedge \frac{n}{m+1} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(dx)^m}{x^m} = 0$

Rule: If $b^2 - 4ac \neq 0 \wedge \frac{n}{m+1} \in \mathbb{Z}$, then

$$\int (dx)^m P_q[x^n] (a+bx^n+cx^{2n})^p dx \rightarrow \frac{(dx)^m}{x^m} \int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d*x_)^m_*Pq_*(a_+b_*x_^n+c_*x_^2n_)^p_,x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

11. $\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx$ when $b^2 - 4ac \neq 0 \wedge p \in \mathbb{Z}^-$

1: $\int \frac{(dx)^m P_q[x]}{a+bx^n+cx^{2n}} dx$ when $b^2 - 4ac \neq 0$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let $q = \sqrt{b^2 - 4ac}$, then $\frac{1}{a+bz+cz^2} = \frac{2c}{q} \frac{1}{b-q+2cz} - \frac{2c}{q} \frac{1}{b+q+2cz}$

■

Rule: If $b^2 - 4ac \neq 0$, let $q = \sqrt{b^2 - 4ac}$, then

$$\int \frac{(dx)^m P_q[x]}{a+bx^n+cx^{2n}} dx \rightarrow \frac{2c}{q} \int \frac{(dx)^m P_q[x]}{b-q+2cx^n} dx - \frac{2c}{q} \int \frac{(dx)^m P_q[x]}{b+q+2cx^n} dx$$

Program code:

```
Int[(d.*x_)^m_.*Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    2*c/q*Int[(d*x)^m*Pq/(b-q+2*c*x^n),x] -
    2*c/q*Int[(d*x)^m*Pq/(b+q+2*c*x^n),x] /;
  FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0]
```

2: $\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx$ when $p \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^-$, then

$$\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \rightarrow \int \text{ExpandIntegrand}[(dx)^m P_q[x] (a+bx^n+cx^{2n})^p, x] dx$$

Program code:

```
Int[(d.*x_)^m_.*Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*Pq*(a+b*x^n+c*x^(2*n))^p,x],x] /;
  FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && ILtQ[p+1,0]
```

X: $\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx$

Rule:

$$\int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx \rightarrow \int (dx)^m P_q[x] (a+bx^n+cx^{2n})^p dx$$

Program code:

```
Int[(d_.**x_)^m_.**Pq_*(a_+b_.**x_^n_+c_.**x_^n2_.)^p_,x_Symbol] :=
  Unintegrable[(d*x)^m**Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
  FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && (PolyQ[Pq,x] || PolyQ[Pq,x^n])
```

S: $\int u^m P_q[v^n] (a+bv^n+cv^{2n})^p dx$ when $v = f + gx \wedge u = hv$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u = hv$, then $\partial_x \frac{u^m}{v^m} = 0$

Rule: If $v = f + gx \wedge u = hv$, then

$$\int u^m P_q[v^n] (a+bv^n+cv^{2n})^p dx \rightarrow \frac{u^m}{g v^m} \text{Subst}\left[\int x^m P_q[x^n] (a+bx^n+cx^{2n})^p dx, x, v\right]$$

Program code:

```
Int[u_^m_.**Pq_*(a_+b_.**v_^n_+c_.**v_^n2_.)^p_,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*SubstFor[v,Pq,x]*(a+b*x^n+c*x^(2*n))^p,x],x,v] /;
  FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && LinearPairQ[u,v,x] && PolyQ[Pq,v^n]
```