Rules for integrands of the form $u\ (a + b\,\mathrm{ArcSech}[c\,x])^n$

1. $\int (a + b\,\mathrm{ArcSech}[c\,x])^n\,dx$ when $n \in \mathbb{Z}^+$

   1. $\int \mathrm{ArcSech}[c\,x]\,dx$

      1: $\int \mathrm{ArcSech}[c\,x]\,dx$

Reference: CRC 591, A&S 4.6.47

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x \mathrm{ArcSech}[c\,x] == -\frac{\sqrt{1+c\,x}\,\sqrt{\frac{1}{1+c\,x}}}{x\,\sqrt{1-c^2\,x^2}}$

Basis: $\partial_x \left(\sqrt{1+c\,x}\,\sqrt{\frac{1}{1+c\,x}}\right) == 0$

Rule:

$$\int \mathrm{ArcSech}[c\,x]\,dx \longrightarrow x\,\mathrm{ArcSech}[c\,x] + \sqrt{1+c\,x}\,\sqrt{\frac{1}{1+c\,x}}\int \frac{1}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[ArcSech[c_.*x_],x_Symbol] :=
  x*ArcSech[c*x] + Sqrt[1+c*x]*Sqrt[1/(1+c*x)]*Int[1/Sqrt[1-c^2*x^2],x] /;
FreeQ[c,x]
```

2: $\int \text{ArcCsch}[c\, x]\, dx$

Reference: CRC 594, A&S 4.6.46

Derivation: Integration by parts

Rule:

$$\int \text{ArcCsch}[c\, x]\, dx \longrightarrow x\, \text{ArcCsch}[c\, x] + \frac{1}{c} \int \frac{1}{x \sqrt{1 + \frac{1}{c^2 x^2}}}\, dx$$

Program code:

```
Int[ArcCsch[c_.*x_],x_Symbol] :=
  x*ArcCsch[c*x] + 1/c*Int[1/(x*Sqrt[1+1/(c^2*x^2)]),x] /;
FreeQ[c,x]
```

2: $\int (a+b\,\text{ArcSech}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $1 = -\frac{1}{c}\,\text{Sech}[\text{ArcSech}[c\,x]]\,\text{Tanh}[\text{ArcSech}[c\,x]]\,\partial_x\text{ArcSech}[c\,x]$

Rule: If $n\in\mathbb{Z}^+$, then

$$\int (a+b\,\text{ArcSech}[c\,x])^n\,dx \longrightarrow -\frac{1}{c}\,\text{Subst}\left[\int (a+b\,x)^n\,\text{Sech}[x]\,\text{Tanh}[x]\,dx,\ x,\ \text{ArcSech}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcSech[c_.*x_])^n_,x_Symbol] :=
  -1/c*Subst[Int[(a+b*x)^n*Sech[x]*Tanh[x],x],x,ArcSech[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCsch[c_.*x_])^n_,x_Symbol] :=
  -1/c*Subst[Int[(a+b*x)^n*Csch[x]*Coth[x],x],x,ArcCsch[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[n,0]
```

2. $\int (d\,x)^m\,(a+b\,\text{ArcSech}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

1. $\int (d\,x)^m\,(a+b\,\text{ArcSech}[c\,x])\,dx$

1: $\int \frac{a+b\,\text{ArcSech}[c\,x]}{x}\,dx$

Derivation: Integration by substitution

Basis: $\text{ArcSech}[z] == \text{ArcCosh}\left[\frac{1}{z}\right]$

Basis: $\frac{F\left[\frac{1}{x}\right]}{x} == -\text{Subst}\left[\frac{F[x]}{x},\,x,\,\frac{1}{x}\right]\,\partial_x\,\frac{1}{x}$

Rule:

$$\int \frac{a+b\,\text{ArcSech}[c\,x]}{x}\,dx\;\longrightarrow\;\int \frac{a+b\,\text{ArcCosh}\left[\frac{1}{c\,x}\right]}{x}\,dx\;\longrightarrow\;-\text{Subst}\left[\int \frac{a+b\,\text{ArcCosh}\left[\frac{x}{c}\right]}{x}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(a_.+b_.*ArcSech[c_.*x_])/x_,x_Symbol] :=
  -Subst[Int[(a+b*ArcCosh[x/c])/x,x],x,1/x] /;
FreeQ[{a,b,c},x]
```

```
Int[(a_.+b_.*ArcCsch[c_.*x_])/x_,x_Symbol] :=
  -Subst[Int[(a+b*ArcSinh[x/c])/x,x],x,1/x] /;
FreeQ[{a,b,c},x]
```

2. $\int (d\,x)^m\,(a+b\,\text{ArcSech}[c\,x])\,dx$ when $m\neq -1$

1: $\int (d\,x)^m\,(a+b\,\text{ArcSech}[c\,x])\,dx$ when $m\neq -1$

Reference: CRC 593', A&S 4.6.58'

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x\,(a+b\,\text{ArcSech}[c\,x]) == -\frac{b\sqrt{\frac{1}{1+c\,x}}}{x\sqrt{1-c\,x}}$

Basis: $\partial_x\left(\sqrt{1+c\,x}\,\sqrt{\frac{1}{1+c\,x}}\right) == 0$

- Note: Although $\sqrt{1-c^2\,x^2} == \sqrt{1-c\,x}\,\sqrt{1+c\,x}$, leaving denominator factored allows for more cancellation with piecewise constant factor.

Rule: If $m\neq -1$, then

$$\int (d\,x)^m\,(a+b\,\text{ArcSech}[c\,x])\,dx \to \frac{(d\,x)^{m+1}\,(a+b\,\text{ArcSech}[c\,x])}{d\,(m+1)}+\frac{b\sqrt{1+c\,x}}{m+1}\sqrt{\frac{1}{1+c\,x}}\int\frac{(d\,x)^m}{\sqrt{1-c\,x}\,\sqrt{1+c\,x}}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcSech[c_.*x_]),x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcSech[c*x])/(d*(m+1)) +
  b*Sqrt[1+c*x]/(m+1)*Sqrt[1/(1+c*x)]*Int[(d*x)^m/(Sqrt[1-c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

2: $\int (d\,x)^m\,(a+b\,\text{ArcCsch}[c\,x])\,dx$ when $m\neq -1$

Reference: CRC 596, A&S 4.6.56

Derivation: Integration by parts

Rule: If $m\neq -1$, then

$$\int (d\,x)^m\,(a+b\,\text{ArcCsch}[c\,x])\,dx \longrightarrow \frac{(d\,x)^{m+1}\,(a+b\,\text{ArcCsch}[c\,x])}{d\,(m+1)}+\frac{b\,d}{c\,(m+1)}\int\frac{(d\,x)^{m-1}}{\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCsch[c_.*x_]),x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcCsch[c*x])/(d*(m+1)) +
  b*d/(c*(m+1))*Int[(d*x)^(m-1)/Sqrt[1+1/(c^2*x^2)],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

2: $\int x^m (a + b \,\text{ArcSech}[c\,x])^n \, dx$ when $n \in \mathbb{Z} \land m \in \mathbb{Z} \land (n > 0 \lor m < -1)$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}$, then

$$x^m F[\text{ArcSech}[c\,x]] = -\frac{1}{c^{m+1}} \text{Subst}\left[F[x] \,\text{Sech}[x]^{m+1} \,\text{Tanh}[x],\, x,\, \text{ArcSech}[c\,x]\right] \partial_x \text{ArcSech}[c\,x]$$

Rule: If $n \in \mathbb{Z} \land m \in \mathbb{Z} \land (n > 0 \lor m < -1)$, then

$$\int x^m (a + b \,\text{ArcSech}[c\,x])^n \, dx \longrightarrow -\frac{1}{c^{m+1}} \text{Subst}\left[\int (a + b\,x)^n \,\text{Sech}[x]^{m+1} \,\text{Tanh}[x] \, dx,\, x,\, \text{ArcSech}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSech[c_.*x_])^n_,x_Symbol] :=
  -1/c^(m+1)*Subst[Int[(a+b*x)^n*Sech[x]^(m+1)*Tanh[x],x],x,ArcSech[c*x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n] && IntegerQ[m] && (GtQ[n,0] || LtQ[m,-1])
```

```
Int[x_^m_.*(a_.+b_.*ArcCsch[c_.*x_])^n_,x_Symbol] :=
  -1/c^(m+1)*Subst[Int[(a+b*x)^n*Csch[x]^(m+1)*Coth[x],x],x,ArcCsch[c*x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n] && IntegerQ[m] && (GtQ[n,0] || LtQ[m,-1])
```

3. $\int (d+e x)^m (a+b \operatorname{ArcSech}[c x])\, dx$

1. $\int (d+e x)^m (a+b \operatorname{ArcSech}[c x])\, dx$

1: $\int \frac{a+b \operatorname{ArcSech}[c x]}{d+e x}\, dx$

Derivation: Integration by parts

- Basis: $\frac{1}{d+e x} == \frac{1}{e}\, \partial_x \left(\operatorname{Log}\left[1+\frac{e-\sqrt{-c^2 d^2+e^2}}{c\, d\, e^{\operatorname{ArcSech}[c x]}}\right]+\operatorname{Log}\left[1+\frac{e+\sqrt{-c^2 d^2+e^2}}{c\, d\, e^{\operatorname{ArcSech}[c x]}}\right]-\operatorname{Log}\left[1+\frac{1}{e^{2 \operatorname{ArcSech}[c x]}}\right]\right)$

- Basis: $\partial_x (a+b \operatorname{ArcSech}[c x]) == -\frac{b \sqrt{\frac{1-c x}{1+c x}}}{x (1-c x)}$

Rule:

$$\int \frac{a+b \operatorname{ArcSech}[c x]}{d+e x}\, dx \longrightarrow$$

$$\frac{(a+b \operatorname{ArcSech}[c x]) \operatorname{Log}\left[1+\frac{e-\sqrt{-c^2 d^2+e^2}}{c\, d\, e^{\operatorname{ArcSech}[c x]}}\right]}{e}+\frac{(a+b \operatorname{ArcSech}[c x]) \operatorname{Log}\left[1+\frac{e+\sqrt{-c^2 d^2+e^2}}{c\, d\, e^{\operatorname{ArcSech}[c x]}}\right]}{e}-\frac{(a+b \operatorname{ArcSech}[c x]) \operatorname{Log}\left[1+\frac{1}{e^{2 \operatorname{ArcSech}[c x]}}\right]}{e}+$$

$$\frac{b}{e} \int \frac{\sqrt{\frac{1-c x}{1+c x}} \operatorname{Log}\left[1+\frac{e-\sqrt{-c^2 d^2+e^2}}{c\, d\, e^{\operatorname{ArcSech}[c x]}}\right]}{x (1-c x)}\, dx+\frac{b}{e} \int \frac{\sqrt{\frac{1-c x}{1+c x}} \operatorname{Log}\left[1+\frac{e+\sqrt{-c^2 d^2+e^2}}{c\, d\, e^{\operatorname{ArcSech}[c x]}}\right]}{x (1-c x)}\, dx-\frac{b}{e} \int \frac{\sqrt{\frac{1-c x}{1+c x}} \operatorname{Log}\left[1+\frac{1}{e^{2 \operatorname{ArcSech}[c x]}}\right]}{x (1-c x)}\, dx$$

Program code:

```
Int[(a_.+b_.*ArcSech[c_.*x_])/(d_.+e_.*x_),x_Symbol] :=
  (a+b*ArcSech[c*x])*Log[1+(e-Sqrt[-c^2*d^2+e^2])/(c*d*E^ArcSech[c*x])]/e +
  (a+b*ArcSech[c*x])*Log[1+(e+Sqrt[-c^2*d^2+e^2])/(c*d*E^ArcSech[c*x])]/e -
  (a+b*ArcSech[c*x])*Log[1+1/E^(2*ArcSech[c*x])]/e +
  b/e*Int[(Sqrt[(1-c*x)/(1+c*x)]*Log[1+(e-Sqrt[-c^2*d^2+e^2])/(c*d*E^ArcSech[c*x])])/(x*(1-c*x)),x] +
  b/e*Int[(Sqrt[(1-c*x)/(1+c*x)]*Log[1+(e+Sqrt[-c^2*d^2+e^2])/(c*d*E^ArcSech[c*x])])/(x*(1-c*x)),x] -
  b/e*Int[(Sqrt[(1-c*x)/(1+c*x)]*Log[1+1/E^(2*ArcSech[c*x])])/(x*(1-c*x)),x] /;
FreeQ[{a,b,c,d,e},x]
```

2: $\int (d+e\,x)^m\,(a+b\,\text{ArcSech}[c\,x])\,dx$ when $m\neq -1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x\,(a+b\,\text{ArcSech}[c\,x]) == -\frac{b\sqrt{1+c\,x}\,\sqrt{\frac{1}{1+c\,x}}}{x\sqrt{1-c^2\,x^2}}$

Basis: $\partial_x\left(\sqrt{1+c\,x}\,\sqrt{\frac{1}{1+c\,x}}\right) == 0$

Rule: If $m\in\mathbb{Z}\land m\neq -1$, then

$$\int (d+e\,x)^m\,(a+b\,\text{ArcSech}[c\,x])\,dx \to \frac{(d+e\,x)^{m+1}\,(a+b\,\text{ArcSech}[c\,x])}{e\,(m+1)}+\frac{b\sqrt{1+c\,x}}{e\,(m+1)}\sqrt{\frac{1}{1+c\,x}}\int\frac{(d+e\,x)^{m+1}}{x\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcSech[c_.*x_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcSech[c*x])/(e*(m+1)) +
  b*Sqrt[1+c*x]/(e*(m+1))*Sqrt[1/(1+c*x)]*Int[(d+e*x)^(m+1)/(x*Sqrt[1-c^2*x^2]),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[m,-1]
```

2. $\int (d+e\,x)^m\,(a+b\,\text{ArcCsch}[c\,x])\,dx$

1: $\int \frac{a+b\,\text{ArcCsch}[c\,x]}{d+e\,x}\,dx$

Derivation: Integration by parts

Basis: $\frac{1}{d+e\,x} = \frac{1}{e}\,\partial_x\left(\text{Log}\left[1-\frac{\left(e-\sqrt{c^2\,d^2+e^2}\right)\,e^{\text{ArcCsch}[c\,x]}}{c\,d}\right]+\text{Log}\left[1-\frac{\left(e+\sqrt{c^2\,d^2+e^2}\right)\,e^{\text{ArcCsch}[c\,x]}}{c\,d}\right]-\text{Log}\left[1-e^{2\,\text{ArcCsch}[c\,x]}\right]\right)$

Basis: $\partial_x\,(a+b\,\text{ArcCsch}[c\,x]) = -\frac{b}{c\,x^2\,\sqrt{1+\frac{1}{c^2\,x^2}}}$

Rule:

$$\int \frac{a+b\,\text{ArcCsch}[c\,x]}{d+e\,x}\,dx \longrightarrow$$

$$\frac{(a+b\,\text{ArcCsch}[c\,x])\,\text{Log}\left[1-\frac{\left(e-\sqrt{c^2\,d^2+e^2}\right)\,e^{\text{ArcCsch}[c\,x]}}{c\,d}\right]}{e}+\frac{(a+b\,\text{ArcCsch}[c\,x])\,\text{Log}\left[1-\frac{\left(e+\sqrt{c^2\,d^2+e^2}\right)\,e^{\text{ArcCsch}[c\,x]}}{c\,d}\right]}{e}-$$

$$\frac{(a+b\,\text{ArcCsch}[c\,x])\,\text{Log}\left[1-e^{2\,\text{ArcCsch}[c\,x]}\right]}{e}+\frac{b}{c\,e}\int\frac{\text{Log}\left[1-\frac{\left(e-\sqrt{c^2\,d^2+e^2}\right)\,e^{\text{ArcCsch}[c\,x]}}{c\,d}\right]}{x^2\,\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx+$$

$$\frac{b}{c\,e}\int\frac{\text{Log}\left[1-\frac{\left(e+\sqrt{c^2\,d^2+e^2}\right)\,e^{\text{ArcCsch}[c\,x]}}{c\,d}\right]}{x^2\,\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx-\frac{b}{c\,e}\int\frac{\text{Log}\left[1-e^{2\,\text{ArcCsch}[c\,x]}\right]}{x^2\,\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcCsch[c_.*x_])/(d_.+e_.*x_),x_Symbol] :=
  (a+b*ArcCsch[c*x])*Log[1-(e-Sqrt[c^2*d^2+e^2])*E^ArcCsch[c*x]/(c*d)]/e +
  (a+b*ArcCsch[c*x])*Log[1-(e+Sqrt[c^2*d^2+e^2])*E^ArcCsch[c*x]/(c*d)]/e -
  (a+b*ArcCsch[c*x])*Log[1-E^(2*ArcCsch[c*x])]/e +
  b/(c*e)*Int[Log[1-(e-Sqrt[c^2*d^2+e^2])*E^ArcCsch[c*x]/(c*d)]/(x^2*Sqrt[1+1/(c^2*x^2)]),x] +
  b/(c*e)*Int[Log[1-(e+Sqrt[c^2*d^2+e^2])*E^ArcCsch[c*x]/(c*d)]/(x^2*Sqrt[1+1/(c^2*x^2)]),x] -
  b/(c*e)*Int[Log[1-E^(2*ArcCsch[c*x])]/(x^2*Sqrt[1+1/(c^2*x^2)]),x] /;
FreeQ[{a,b,c,d,e},x]
```

2: $\int (d+e\,x)^m\,(a+b\,\text{ArcCsch}[c\,x])\,dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{ArcCsch}[c\,x]) == -\frac{b}{c\,x^2\sqrt{1+\frac{1}{c^2\,x^2}}}$

Rule: If $m \neq -1$, then

$$\int (d+e\,x)^m\,(a+b\,\text{ArcCsch}[c\,x])\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,(a+b\,\text{ArcCsch}[c\,x])}{e\,(m+1)}+\frac{b}{c\,e\,(m+1)}\int \frac{(d+e\,x)^{m+1}}{x^2\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCsch[c_.*x_]),x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcCsch[c*x])/(e*(m+1)) +
  b/(c*e*(m+1))*Int[(d+e*x)^(m+1)/(x^2*Sqrt[1+1/(c^2*x^2)]),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[m,-1]
```

4. $\int (d+e x^2)^p (a+b \operatorname{ArcSech}[c x])^n \, dx$ when $n \in \mathbb{Z}^+$

1. $\int (d+e x^2)^p (a+b \operatorname{ArcSech}[c x]) \, dx$ when $p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-$

1: $\int (d+e x^2)^p (a+b \operatorname{ArcSech}[c x]) \, dx$ when $p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x (a+b \operatorname{ArcSech}[c x]) == -\frac{b \sqrt{\frac{1}{1+c x}}}{x \sqrt{1-c x}}$

Basis: $\partial_x \left(\sqrt{\frac{1}{1+c x}} \sqrt{1+c x}\right) == 0$

Note: If $p+\frac{1}{2} \in \mathbb{Z}$, then the terms of $\int (d+e x^2)^p \, dx$ times $\partial_x (a+b \operatorname{ArcSech}[c x])$ are of an easily integrable form.

Rule: If $p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-$, let $u = \int (d+e x^2)^p \, dx$, then

$$\int (d+e x^2)^p (a+b \operatorname{ArcSech}[c x]) \, dx \to u (a+b \operatorname{ArcSech}[c x]) + b \sqrt{1+c x} \sqrt{\frac{1}{1+c x}} \int \frac{u}{x \sqrt{1-c x} \sqrt{1+c x}} \, dx$$

Program code:

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSech[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[(a+b*ArcSech[c*x]),u,x] + b*Sqrt[1+c*x]*Sqrt[1/(1+c*x)]*Int[SimplifyIntegrand[u/(x*Sqrt[1-c*x]*Sqrt[1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

2: $\int (d+e x^2)^p (a+b \operatorname{ArcCsch}[c x]) \, dx$ when $p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x\,(a+b\,\text{ArcCsch}[c\,x]) == \frac{b\,c}{\sqrt{-c^2 x^2}\,\sqrt{-1-c^2 x^2}}$

Basis: $\partial_x \frac{x}{\sqrt{-c^2 x^2}} == 0$

Note: If $p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-$, then $\int (d+e\,x^2)^p\,dx$ is expressible as an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.

Rule: If $p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-$, let $u = \int (d+e\,x^2)^p\,dx$, then

$$\int (d+e\,x^2)^p\,(a+b\,\text{ArcCsch}[c\,x])\,dx \to u\,(a+b\,\text{ArcCsch}[c\,x]) - b\,c\int \frac{u}{\sqrt{-c^2 x^2}\,\sqrt{-1-c^2 x^2}}\,dx \to u\,(a+b\,\text{ArcCsch}[c\,x]) - \frac{b\,c\,x}{\sqrt{-c^2 x^2}}\int \frac{u}{x\,\sqrt{-1-c^2 x^2}}\,dx$$

Program code:

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsch[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[(a+b*ArcCsch[c*x]),u,x] - b*c*x/Sqrt[-c^2*x^2]*Int[SimplifyIntegrand[u/(x*Sqrt[-1-c^2*x^2]),x],x]] /;
FreeQ[{a,b,c,d,e},x] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

2: $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSech}[c\,x])^n\,dx$ when $n \in \mathbb{Z}^+ \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{ArcSech}[z] == \text{ArcCosh}\left[\frac{1}{z}\right]$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x \frac{1}{x}$

Rule: If $n \in \mathbb{Z}^+ \land p \in \mathbb{Z}$, then

$$\int (d+e\,x^2)^p\,(a+b\,\text{ArcSech}[c\,x])^n\,dx \to \int \left(\frac{1}{x}\right)^{-2p}\left(e+\frac{d}{x^2}\right)^p\left(a+b\,\text{ArcCosh}\left[\frac{1}{c\,x}\right]\right)^n dx$$

$$\to -\text{Subst}\left[\int \frac{\left(e+d\,x^2\right)^p \left(a+b\,\text{ArcCosh}\left[\frac{x}{c}\right]\right)^n}{x^{2\,(p+1)}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSech[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcCosh[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegerQ[p]
```

```
Int[(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsch[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcSinh[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegerQ[p]
```

3. $\int \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSech}[c\,x]\right)^n dx$ when $n \in \mathbb{Z}^+ \land c^2 d+e == 0 \land p+\frac{1}{2} \in \mathbb{Z}$

1: $\int \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSech}[c\,x]\right)^n dx$ when $n \in \mathbb{Z}^+ \land c^2 d+e == 0 \land p+\frac{1}{2} \in \mathbb{Z} \land e>0 \land d<0$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{d+e\,x^2}}{x\sqrt{e+\frac{d}{x^2}}} == 0$

  Basis: $\text{ArcSech}[z] == \text{ArcCosh}\left[\frac{1}{z}\right]$

  Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2},\ x,\ \frac{1}{x}\right] \partial_x \frac{1}{x}$

- Basis: If $e>0 \land d<0$, then $\frac{\sqrt{d+e\,x^2}}{\sqrt{e+\frac{d}{x^2}}} == \sqrt{x^2}$

Rule: If $n \in \mathbb{Z}^+ \land c^2 d+e == 0 \land p+\frac{1}{2} \in \mathbb{Z} \land e>0 \land d<0$, then

$$\int (d+e x^2)^p (a+b\,\text{ArcSech}[c x])^n \,dx \rightarrow \frac{\sqrt{d+e x^2}}{x\sqrt{e+\frac{d}{x^2}}}\int \left(\frac{1}{x}\right)^{-2p}\left(e+\frac{d}{x^2}\right)^p \left(a+b\,\text{ArcCosh}\left[\frac{1}{c x}\right]\right)^n dx$$

$$\rightarrow -\frac{\sqrt{x^2}}{x}\,\text{Subst}\left[\int \frac{(e+d x^2)^p \left(a+b\,\text{ArcCosh}\left[\frac{x}{c}\right]\right)^n}{x^{2(p+1)}}dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSech[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcCosh[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsch[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcSinh[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[e-c^2*d,0] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

2: $\int (d+e x^2)^p (a+b\,\text{ArcSech}[c x])^n \,dx$ when $n \in \mathbb{Z}^+ \land c^2 d+e == 0 \land p+\frac{1}{2} \in \mathbb{Z} \land \neg (e>0 \land d<0)$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{d+e x^2}}{x\sqrt{e+\frac{d}{x^2}}} == 0$

  Basis: $\text{ArcSech}[z] == \text{ArcCosh}\left[\frac{1}{z}\right]$

  Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2},\ x,\ \frac{1}{x}\right]\partial_x \frac{1}{x}$

  Rule: If $c^2 d+e == 0 \land p+\frac{1}{2} \in \mathbb{Z} \land \neg (e>0 \land d<0)$, then

$$\int (d+e x^2)^p (a+b\,\text{ArcSech}[c x])^n \,dx \rightarrow \frac{\sqrt{d+e x^2}}{x\sqrt{e+\frac{d}{x^2}}}\int \left(\frac{1}{x}\right)^{-2p}\left(e+\frac{d}{x^2}\right)^p \left(a+b\,\text{ArcCosh}\left[\frac{1}{c x}\right]\right)^n dx$$

$$\to -\frac{\sqrt{d+e\,x^2}}{x\,\sqrt{e+\frac{d}{x^2}}}\,\text{Subst}\left[\int\frac{\left(e+d\,x^2\right)^p\,\left(a+b\,\text{ArcCosh}\left[\frac{x}{c}\right]\right)^n}{x^{2\,(p+1)}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSech[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcCosh[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

```
Int[(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsch[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcSinh[x/c])^n/x^(2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[e-c^2*d,0] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

5. $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSech}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

   1. $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSech}[c\,x])\,dx$ when

      $\left(p\in\mathbb{Z}^+\land\neg\left(\frac{m-1}{2}\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor\left(\frac{m+1}{2}\in\mathbb{Z}^+\land\neg\left(p\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor\left(\frac{m+2p+1}{2}\in\mathbb{Z}^-\land\frac{m-1}{2}\notin\mathbb{Z}^-\right)$

      1. $\int x\left(d+e\,x^2\right)^p (a+b\,\text{ArcSech}[c\,x])\,dx$ when $p\neq-1$

         1: $\int x\left(d+e\,x^2\right)^p (a+b\,\text{ArcSech}[c\,x])\,dx$ when $p\neq-1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $x\left(d+e\,x^2\right)^p == \partial_x \frac{\left(d+e\,x^2\right)^{p+1}}{2\,e\,(p+1)}$

Basis: $\partial_x\,(a+b\,\text{ArcSech}[c\,x]) == -\frac{b\sqrt{\frac{1}{1+c\,x}}}{x\sqrt{1-c\,x}}$

Basis: $\partial_x\left(\sqrt{1+c\,x}\,\sqrt{\frac{1}{1+c\,x}}\right) == 0$

- Note: Although $\sqrt{1-c^2\,x^2} == \sqrt{1-c\,x}\,\sqrt{1+c\,x}$, leaving denominator factored allows for more cancellation with piecewise constant factor.

Rule: If $p\neq-1$, then

$$\int x\left(d+e\,x^2\right)^p (a+b\,\text{ArcSech}[c\,x])\,dx \to \frac{\left(d+e\,x^2\right)^{p+1}(a+b\,\text{ArcSech}[c\,x])}{2\,e\,(p+1)}+\frac{b\sqrt{1+c\,x}}{2\,e\,(p+1)}\sqrt{\frac{1}{1+c\,x}}\int\frac{\left(d+e\,x^2\right)^{p+1}}{x\sqrt{1-c\,x}\,\sqrt{1+c\,x}}\,dx$$

Program code:

```
Int[x_*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSech[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcSech[c*x])/(2*e*(p+1)) +
  b*Sqrt[1+c*x]/(2*e*(p+1))*Sqrt[1/(1+c*x)]*Int[(d+e*x^2)^(p+1)/(x*Sqrt[1-c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[p,-1]
```

2: $\int x\left(d+e\,x^2\right)^p\left(a+b\,\text{ArcCsch}[c\,x]\right)dx$ when $p\neq-1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $x\left(d+e\,x^2\right)^p==\partial_x\frac{\left(d+e\,x^2\right)^{p+1}}{2\,e\,(p+1)}$

Basis: $\partial_x\left(a+b\,\text{ArcCsch}[c\,x]\right)==\frac{b\,c}{\sqrt{-c^2\,x^2}\,\sqrt{-1-c^2\,x^2}}$

Basis: $\partial_x\frac{x}{\sqrt{-c^2\,x^2}}==0$

Rule: If $p\neq-1$, then

$$\int x\left(d+e\,x^2\right)^p\left(a+b\,\text{ArcCsch}[c\,x]\right)dx\to\frac{\left(d+e\,x^2\right)^{p+1}\left(a+b\,\text{ArcCsch}[c\,x]\right)}{2\,e\,(p+1)}-\frac{b\,c}{2\,e\,(p+1)}\int\frac{\left(d+e\,x^2\right)^{p+1}}{\sqrt{-c^2\,x^2}\,\sqrt{-1-c^2\,x^2}}dx$$

$$\to\frac{\left(d+e\,x^2\right)^{p+1}\left(a+b\,\text{ArcCsch}[c\,x]\right)}{2\,e\,(p+1)}-\frac{b\,c\,x}{2\,e\,(p+1)\,\sqrt{-c^2\,x^2}}\int\frac{\left(d+e\,x^2\right)^{p+1}}{x\,\sqrt{-1-c^2\,x^2}}dx$$

Program code:

```
Int[x_*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsch[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcCsch[c*x])/(2*e*(p+1)) -
  b*c*x/(2*e*(p+1)*Sqrt[-c^2*x^2])*Int[(d+e*x^2)^(p+1)/(x*Sqrt[-1-c^2*x^2]),x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[p,-1]
```

2. $\int (f x)^m (d+e x^2)^p (a+b \text{ArcSech}[c x]) \, dx$ when
$\left(p \in \mathbb{Z}^+ \land \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \land m+2p+3>0\right)\right) \lor \left(\frac{m+1}{2} \in \mathbb{Z}^+ \land \neg (p \in \mathbb{Z}^- \land m+2p+3>0)\right) \lor \left(\frac{m+2p+1}{2} \in \mathbb{Z}^- \land \frac{m-1}{2} \notin \mathbb{Z}^-\right)$

1: $\int (f x)^m (d+e x^2)^p (a+b \text{ArcSech}[c x]) \, dx$ when
$\left(p \in \mathbb{Z}^+ \land \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \land m+2p+3>0\right)\right) \lor \left(\frac{m+1}{2} \in \mathbb{Z}^+ \land \neg (p \in \mathbb{Z}^- \land m+2p+3>0)\right) \lor \left(\frac{m+2p+1}{2} \in \mathbb{Z}^- \land \frac{m-1}{2} \notin \mathbb{Z}^-\right)$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x (a+b \text{ArcSech}[c x]) = -\frac{b \sqrt{\frac{1}{1+c x}}}{x \sqrt{1-c x}}$

Basis: $\partial_x \left(\sqrt{1+c x} \sqrt{\frac{1}{1+c x}}\right) = 0$

Note: Although $\sqrt{1-c^2 x^2} = \sqrt{1-c x} \sqrt{1+c x}$, leaving denominator factored allows for more cancellation with piecewise constant factor.

Note: If $\left(p \in \mathbb{Z}^+ \land \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \land m+2p+3>0\right)\right) \lor$,
$\left(\frac{m+1}{2} \in \mathbb{Z}^+ \land \neg (p \in \mathbb{Z}^- \land m+2p+3>0)\right) \lor \left(\frac{m+2p+1}{2} \in \mathbb{Z}^- \land \frac{m-1}{2} \notin \mathbb{Z}^-\right)$
then $\int (f x)^m (d+e x^2)^p \, dx$ is expressible as an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.

Rule: If $\left(p \in \mathbb{Z}^+ \land \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \land m+2p+3>0\right)\right) \lor \left(\frac{m+1}{2} \in \mathbb{Z}^+ \land \neg (p \in \mathbb{Z}^- \land m+2p+3>0)\right) \lor \left(\frac{m+2p+1}{2} \in \mathbb{Z}^- \land \frac{m-1}{2} \notin \mathbb{Z}^-\right)$, let $u = \int (f x)^m (d+e x^2)^p \, dx$, then

$$\int (f x)^m (d+e x^2)^p (a+b \text{ArcSech}[c x]) \, dx \to u (a+b \text{ArcSech}[c x]) + b \sqrt{1+c x} \sqrt{\frac{1}{1+c x}} \int \frac{u}{x \sqrt{1-c x} \sqrt{1+c x}} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSech[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[(a+b*ArcSech[c*x]),u,x] + b*Sqrt[1+c*x]*Sqrt[1/(1+c*x)]*Int[SimplifyIntegrand[u/(x*Sqrt[1-c*x]*Sqrt[1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && (
  IGtQ[p,0] && Not[ILtQ[(m-1)/2,0] && GtQ[m+2*p+3,0]] ||
  IGtQ[(m+1)/2,0] && Not[ILtQ[p,0] && GtQ[m+2*p+3,0]] ||
  ILtQ[(m+2*p+1)/2,0] && Not[ILtQ[(m-1)/2,0]])
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCsch}[c\,x])\,dx$ when
$\left(p\in\mathbb{Z}^+\land\neg\left(\frac{m-1}{2}\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor\left(\frac{m+1}{2}\in\mathbb{Z}^+\land\neg\left(p\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor\left(\frac{m+2p+1}{2}\in\mathbb{Z}^-\land\frac{m-1}{2}\notin\mathbb{Z}^-\right)$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x (a+b\,\text{ArcCsch}[c\,x]) == \frac{b\,c}{\sqrt{-c^2 x^2}\,\sqrt{-1-c^2 x^2}}$

Basis: $\partial_x \frac{x}{\sqrt{-c^2 x^2}} == 0$

Note: If $\left(p\in\mathbb{Z}^+\land\neg\left(\frac{m-1}{2}\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor$,
$\left(\frac{m+1}{2}\in\mathbb{Z}^+\land\neg\left(p\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor\left(\frac{m+2p+1}{2}\in\mathbb{Z}^-\land\frac{m-1}{2}\notin\mathbb{Z}^-\right)$
then $\int (f\,x)^m \left(d+e\,x^2\right)^p dx$ is expressible as an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.

Rule: If $\left(p\in\mathbb{Z}^+\land\neg\left(\frac{m-1}{2}\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor\left(\frac{m+1}{2}\in\mathbb{Z}^+\land\neg\left(p\in\mathbb{Z}^-\land m+2p+3>0\right)\right)\lor\left(\frac{m+2p+1}{2}\in\mathbb{Z}^-\land\frac{m-1}{2}\notin\mathbb{Z}^-\right)$, let $u=\int (f\,x)^m \left(d+e\,x^2\right)^p dx$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCsch}[c\,x])\,dx \to u\,(a+b\,\text{ArcCsch}[c\,x]) -$$
$$b\,c\int\frac{u}{\sqrt{-c^2 x^2}\,\sqrt{-1-c^2 x^2}}\,dx \to u\,(a+b\,\text{ArcCsch}[c\,x]) - \frac{b\,c\,x}{\sqrt{-c^2 x^2}}\int\frac{u}{x\,\sqrt{-1-c^2 x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsch[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[(a+b*ArcCsch[c*x]),u,x] - b*c*x/Sqrt[-c^2*x^2]*Int[SimplifyIntegrand[u/(x*Sqrt[-1-c^2*x^2]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && (
  IGtQ[p,0] && Not[ILtQ[(m-1)/2,0] && GtQ[m+2*p+3,0]] ||
  IGtQ[(m+1)/2,0] && Not[ILtQ[p,0] && GtQ[m+2*p+3,0]] ||
  ILtQ[(m+2*p+1)/2,0] && Not[ILtQ[(m-1)/2,0]] )
```

2: $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSech}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \land (m \mid p) \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\operatorname{ArcSech}[z] == \operatorname{ArcCosh}\left[\frac{1}{z}\right]$

Basis: $F\left[\frac{1}{x}\right] == -\operatorname{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule: If $n \in \mathbb{Z}^+ \land (m \mid p) \in \mathbb{Z}$, then

$$\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSech}[c x])^n \, dx \to \int \left(\frac{1}{x}\right)^{-m-2p} \left(e+\frac{d}{x^2}\right)^p \left(a+b \operatorname{ArcCosh}\left[\frac{1}{c x}\right]\right)^n dx$$

$$\to -\operatorname{Subst}\left[\int \frac{\left(e+d x^2\right)^p \left(a+b \operatorname{ArcCosh}\left[\frac{x}{c}\right]\right)^n}{x^{m+2(p+1)}} \, dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcSech[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcCosh[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegersQ[m,p]
```

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_.*(a_.+b_.*ArcCsch[c_.*x_])^n_.,x_Symbol] :=
  -Subst[Int[(e+d*x^2)^p*(a+b*ArcSinh[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && IntegersQ[m,p]
```

3. $\int x^m \left(d+e x^2\right)^p (a+b\,\text{ArcSech}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\land c^2 d+e=0\land m\in\mathbb{Z}\land p+\frac{1}{2}\in\mathbb{Z}$

1: $\int x^m \left(d+e x^2\right)^p (a+b\,\text{ArcSech}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\land c^2 d+e=0\land m\in\mathbb{Z}\land p+\frac{1}{2}\in\mathbb{Z}\land e>0\land d<0$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{d+e x^2}}{x\sqrt{e+\frac{d}{x^2}}}=0$

Basis: $\text{ArcSech}[z]=\text{ArcCosh}\left[\frac{1}{z}\right]$

Basis: $F\left[\frac{1}{x}\right]=-\text{Subst}\left[\frac{F[x]}{x^2},x,\frac{1}{x}\right]\partial_x \frac{1}{x}$

- Basis: If $e>0\land d<0$, then $\frac{\sqrt{d+e x^2}}{\sqrt{e+\frac{d}{x^2}}}=\sqrt{x^2}$

Rule: If $n\in\mathbb{Z}^+\land c^2 d+e=0\land m\in\mathbb{Z}\land p+\frac{1}{2}\in\mathbb{Z}\land e>0\land d<0$, then

$$\int x^m \left(d+e x^2\right)^p (a+b\,\text{ArcSech}[c\,x])^n\,dx\to\frac{\sqrt{d+e x^2}}{x\sqrt{e+\frac{d}{x^2}}}\int\left(\frac{1}{x}\right)^{-m-2p}\left(e+\frac{d}{x^2}\right)^p\left(a+b\,\text{ArcCosh}\left[\frac{1}{c\,x}\right]\right)^n dx$$

$$\to-\frac{\sqrt{x^2}}{x}\,\text{Subst}\left[\int\frac{\left(e+d x^2\right)^p\left(a+b\,\text{ArcCosh}\left[\frac{x}{c}\right]\right)^n}{x^{m+2(p+1)}}\,dx,x,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSech[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcCosh[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsch[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[x^2]/x*Subst[Int[(e+d*x^2)^p*(a+b*ArcSinh[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[e-c^2*d,0] && IntegerQ[m] && IntegerQ[p+1/2] && GtQ[e,0] && LtQ[d,0]
```

2: $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSech}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \wedge c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge p+\frac{1}{2} \in \mathbb{Z} \wedge \neg (e>0 \wedge d<0)$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} = 0$

Basis: $\operatorname{ArcSech}[z] = \operatorname{ArcCosh}\left[\frac{1}{z}\right]$

Basis: $F\left[\frac{1}{x}\right] = -\operatorname{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule: If $n \in \mathbb{Z}^+ \wedge c^2 d+e=0 \wedge m \in \mathbb{Z} \wedge p+\frac{1}{2} \in \mathbb{Z} \wedge \neg (e>0 \wedge d<0)$, then

$$\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSech}[c x])^n \, dx \to \frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} \int \left(\frac{1}{x}\right)^{-m-2 p} \left(e+\frac{d}{x^2}\right)^p \left(a+b \operatorname{ArcCosh}\left[\frac{1}{c x}\right]\right)^n dx$$

$$\to -\frac{\sqrt{d+e x^2}}{x \sqrt{e+\frac{d}{x^2}}} \operatorname{Subst}\left[\int \frac{\left(e+d x^2\right)^p \left(a+b \operatorname{ArcCosh}\left[\frac{x}{c}\right]\right)^n}{x^{m+2 (p+1)}} dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcSech[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcCosh[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[c^2*d+e,0] && IntegerQ[m] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

```
Int[x_^m_.*(d_.+e_.*x_^2)^p_*(a_.+b_.*ArcCsch[c_.*x_])^n_.,x_Symbol] :=
  -Sqrt[d+e*x^2]/(x*Sqrt[e+d/x^2])*Subst[Int[(e+d*x^2)^p*(a+b*ArcSinh[x/c])^n/x^(m+2*(p+1)),x],x,1/x] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[n,0] && EqQ[e-c^2*d,0] && IntegerQ[m] && IntegerQ[p+1/2] && Not[GtQ[e,0] && LtQ[d,0]]
```

6. $\int u\,(a+b\,\text{ArcSech}[c\,x])\,dx$ when $\int u\,dx$ is free of inverse functions

1: $\int u\,(a+b\,\text{ArcSech}[c\,x])\,dx$ when $\int u\,dx$ is free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x\,(a+b\,\text{ArcSech}[c\,x]) == -\frac{b}{c\,x^2\,\sqrt{-1+\frac{1}{c\,x}}\,\sqrt{1+\frac{1}{c\,x}}}$

Basis: $\partial_x\,\frac{\sqrt{1-c^2\,x^2}}{x\,\sqrt{-1+\frac{1}{c\,x}}\,\sqrt{1+\frac{1}{c\,x}}} == 0$

Rule: Let $v \to \int u\,dx$, if v is free of inverse functions, then

$$\int u\,(a+b\,\text{ArcSech}[c\,x])\,dx \longrightarrow v\,(a+b\,\text{ArcSech}[c\,x])+\frac{b}{c}\int\frac{v}{x^2\,\sqrt{-1+\frac{1}{c\,x}}\,\sqrt{1+\frac{1}{c\,x}}}\,dx$$

$$\longrightarrow v\,(a+b\,\text{ArcSech}[c\,x])+\frac{b\,\sqrt{1-c^2\,x^2}}{c\,x\,\sqrt{-1+\frac{1}{c\,x}}\,\sqrt{1+\frac{1}{c\,x}}}\int\frac{v}{x\,\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[u_*(a_.+b_.*ArcSech[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[(a+b*ArcSech[c*x]),v,x] +
  b*Sqrt[1-c^2*x^2]/(c*x*Sqrt[-1+1/(c*x)]*Sqrt[1+1/(c*x)])*
    Int[SimplifyIntegrand[v/(x*Sqrt[1-c^2*x^2]),x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

2: $\int u\,(a+b\,\text{ArcCsch}[c\,x])\,dx$ when $\int u\,dx$ is free of inverse functions

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{ArcCsch}[c\,x]) == -\frac{b}{c\,x^2\sqrt{1+\frac{1}{c^2\,x^2}}}$

Rule: Let $v == \int u\,dx$, if v is free of inverse functions, then

$$\int u\,(a+b\,\text{ArcCsch}[c\,x])\,dx \longrightarrow v\,(a+b\,\text{ArcCsch}[c\,x])+\frac{b}{c}\int\frac{v}{x^2\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx$$

Program code:

```
Int[u_*(a_.+b_.*ArcCsch[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[(a+b*ArcCsch[c*x]),v,x] +
  b/c*Int[SimplifyIntegrand[v/(x^2*Sqrt[1+1/(c^2*x^2)]),x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

X: $\int u\,(a+b\,\text{ArcSech}[c\,x])^n\,dx$

Rule:

$$\int u\,(a+b\,\text{ArcSech}[c\,x])^n\,dx \longrightarrow \int u\,(a+b\,\text{ArcSech}[c\,x])^n\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*ArcSech[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcSech[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```

```
Int[u_.*(a_.+b_.*ArcCsch[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcCsch[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```