

Rules for integrands of the form $(c + d x)^m (a + b (F^{g(e+f x)})^n)^p$

1. $\int (c + d x)^m (b F^{g(e+f x)})^n dx$

If the control variable `$UseGamma` is `True`, antiderivatives of expressions of the form $(d + e x)^m (F^{c(a+b x)})^n$ will be much more compactly expressed in terms of the `Gamma` function instead of elementary functions.

```
$UseGamma=False;
```

1: $\int (c + d x)^m (b F^{g(e+f x)})^n dx$ when $m > 0 \wedge 2m \in \mathbb{Z}$

Derivation: Integration by parts

Basis: $(b F^{g(e+f x)})^n = \partial_x \frac{(b F^{g(e+f x)})^n}{f g n \log[F]}$

Rule: If $m > 0 \wedge 2m \in \mathbb{Z}$, then

$$\int (c + d x)^m (b F^{g(e+f x)})^n dx \rightarrow \frac{(c + d x)^m (b F^{g(e+f x)})^n}{f g n \log[F]} - \frac{d m}{f g n \log[F]} \int (c + d x)^{m-1} (b F^{g(e+f x)})^n dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(b_.*F^(g_.*(e_.+f_.*x_)))^n_,x_Symbol] :=
  (c+d*x)^m*(b*F^(g*(e+f*x)))^n/(f*g*n*Log[F]) -
  d*m/(f*g*n*Log[F])*Int[(c+d*x)^(m-1)*(b*F^(g*(e+f*x)))^n,x] /;
FreeQ[{F,b,c,d,e,f,g,n},x] && GtQ[m,0] && IntegerQ[2*m] && Not[$UseGamma===True]
```

2: $\int (c+dx)^m (b F^{g(e+fx)})^n dx$ when $m < -1 \wedge 2m \in \mathbb{Z}$

Derivation: Integration by parts

Basis: $(c+dx)^m = \partial_x \frac{(c+dx)^{m+1}}{d(m+1)}$

Rule: If $m < -1 \wedge 2m \in \mathbb{Z}$, then

$$\int (c+dx)^m (b F^{g(e+fx)})^n dx \rightarrow \frac{(c+dx)^{m+1} (b F^{g(e+fx)})^n}{d(m+1)} - \frac{f g n \text{Log}[F]}{d(m+1)} \int (c+dx)^{m+1} (b F^{g(e+fx)})^n dx$$

Program code:

```
Int[(c_.+d_.**x_)^m_*(b_.*F^(g_.*(e_.+f_.**x_)))^n_,x_Symbol] :=
  (c+d*x)^(m+1)*(b*F^(g*(e+f*x)))^n/(d*(m+1)) -
  f*g*n*Log[F]/(d*(m+1))*Int[(c+d*x)^(m+1)*(b*F^(g*(e+f*x)))^n,x] /;
FreeQ[{F,b,c,d,e,f,g,n},x] && LtQ[m,-1] && IntegerQ[2*m] && Not[$UseGamma===True]
```

$$3. \int (c+dx)^m F^{g(e+fx)} dx$$

$$1. \int (c+dx)^m F^{g(e+fx)} dx \text{ when } m \in \mathbb{Z}$$

$$1: \int \frac{F^{g(e+fx)}}{c+dx} dx$$

$$\text{Basis: ExpIntegralEi}'[z] == \frac{e^z}{z}$$

Rule:

$$\int \frac{F^{g(e+fx)}}{c+dx} dx \rightarrow \frac{1}{d} F^{g\left(e-\frac{cf}{d}\right)} \text{ExpIntegralEi}\left[\frac{f g(c+dx) \text{Log}[F]}{d}\right]$$

Program code:

```
Int[F^(g.*(e_.+f_.*x_))/(c_.+d_.*x_),x_Symbol] :=
  F^(g*(e-c*f/d))/d*ExpIntegralEi[f*g*(c+d*x)*Log[F]/d] /;
FreeQ[{F,c,d,e,f,g},x] && Not[$UseGamma===True]
```

2: $\int (c+dx)^m F^{g(e+fx)} dx$ when $m \in \mathbb{Z}$

Rule: If $m \in \mathbb{Z}$, then

$$\int (c+dx)^m F^{g(e+fx)} dx \rightarrow \frac{(-d)^m F^{g\left(e-\frac{cf}{d}\right)}}{f^{m+1} g^{m+1} \text{Log}[F]^{m+1}} \text{Gamma}\left[m+1, -\frac{f g \text{Log}[F]}{d} (c+dx)\right]$$

Program code:

```
Int[(c_.+d_.**x_)^m_.*F^(g_.*(e_.+f_.**x_)),x_Symbol] :=
  (-d)^m * F^(g*(e-c*f/d)) / (f^(m+1) * g^(m+1) * Log[F]^(m+1)) * Gamma[m+1, -f*g*Log[F]/d*(c+d*x)] /;
FreeQ[{F,c,d,e,f,g},x] && IntegerQ[m]
```

2. $\int (c+dx)^m F^g(e+fx) dx$ when $m \notin \mathbb{Z}$

1: $\int \frac{F^g(e+fx)}{\sqrt{c+dx}} dx$

Derivation: Integration by substitution

Basis: $\frac{F^g(e+fx)}{\sqrt{c+dx}} = \frac{2}{d} \text{Subst} \left[F^g \left(e - \frac{cf}{d} + \frac{fgx^2}{d} \right), x, \sqrt{c+dx} \right] \partial_x \sqrt{c+dx}$

Rule:

$$\int \frac{F^g(e+fx)}{\sqrt{c+dx}} dx \rightarrow \frac{2}{d} \text{Subst} \left[\int F^g \left(e - \frac{cf}{d} + \frac{fgx^2}{d} \right) dx, x, \sqrt{c+dx} \right]$$

Program code:

```
Int[F^(g.*(e_.+f_.**x_))/Sqrt[c_.+d_.**x_],x_Symbol] :=
  2/d*Subst[Int[F^(g*(e-c*f/d)+f*g*x^2/d),x],x,Sqrt[c+d*x]] /;
FreeQ[{F,c,d,e,f,g},x] && Not[$UseGamma===True]
```

2: $\int (c+dx)^m F^{g(e+fx)} dx$ when $m \notin \mathbb{Z}$

Rule: If $2m \notin \mathbb{Z}$, then

$$\int (c+dx)^m F^{g(e+fx)} dx \rightarrow -\frac{F^{g\left(e-\frac{cf}{d}\right)} (c+dx)^{\text{FracPart}[m]}}{d \left(-\frac{fg \text{Log}[F]}{d}\right)^{\text{IntPart}[m]+1} \left(-\frac{fg \text{Log}[F] (c+dx)}{d}\right)^{\text{FracPart}[m]}} \text{Gamma}\left[m+1, -\frac{fg \text{Log}[F]}{d} (c+dx)\right]$$

Program code:

```
Int[(c_.+d_.**x_)^m_*F^(g_.*(e_.+f_.**x_)),x_Symbol] :=
  -F^(g*(e-c*f/d))*(c+d*x)^FracPart[m]/(d*(-f*g*Log[F]/d)^(IntPart[m]+1)*(-f*g*Log[F]*(c+d*x)/d)^FracPart[m])*
  Gamma[m+1,(-f*g*Log[F]/d)*(c+d*x)] /;
FreeQ[{F,c,d,e,f,g,m},x] && Not[IntegerQ[m]]
```

4: $\int (c+dx)^m (b F^g(e+fx))^n dx$

Derivation: Piecewise constant extraction

■ Basis: $\partial_x \frac{(b F^g(e+fx))^n}{F^{gn}(e+fx)} == 0$

Rule:

$$\int (c+dx)^m (b F^g(e+fx))^n dx \rightarrow \frac{(b F^g(e+fx))^n}{F^{gn}(e+fx)} \int (c+dx)^m F^{gn}(e+fx) dx$$

Program code:

```
Int[(c_.+d_.**x_)^m_.*(b_.*F^(g_.*(e_.+f_.**x_)))^n_,x_Symbol] :=
  (b.*F^(g*(e+f*x)))^n/F^(g*n*(e+f*x))*Int[(c+d*x)^m.*F^(g*n*(e+f*x)),x] /;
FreeQ[{F,b,c,d,e,f,g,m,n},x]
```

2: $\int (c+dx)^m (a+b(F^g(e+fx))^n)^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (c+dx)^m (a+b(F^g(e+fx))^n)^p dx \rightarrow \int (c+dx)^m \text{ExpandIntegrand}[(a+b(F^g(e+fx))^n)^p, x] dx$$

Program code:

```
Int[(c_.+d_.**x_)^m_.*(a_.+b_.*(F^(g_.*(e_.+f_.**x_)))^n_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m, (a+b*(F^(g*(e+f*x)))^n)^p,x],x] /;
FreeQ[{F,a,b,c,d,e,f,g,m,n},x] && IGtQ[p,0]
```

3: $\int \frac{(c+dx)^m}{a+b(F^g(e+fx))^n} dx$ when $m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+bz} = \frac{1}{a} - \frac{bz}{a(a+bz)}$

Rule: If $m \in \mathbb{Z}^+$, then

$$\int \frac{(c+dx)^m}{a+b(F^g(e+fx))^n} dx \rightarrow \frac{(c+dx)^{m+1}}{a d (m+1)} - \frac{b}{a} \int \frac{(c+dx)^m (F^g(e+fx))^n}{a+b(F^g(e+fx))^n} dx$$

Program code:

```
Int[(c_.+d_.**x_)^m_./(a_.+b_.*(F^(g_.*(e_.+f_.**x_)))^n_.),x_Symbol] :=
  (c+d*x)^(m+1)/(a*d*(m+1)) - b/a*Int[(c+d*x)^m*(F^(g*(e+f*x)))^n/(a+b*(F^(g*(e+f*x)))^n),x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && IGtQ[m,0]
```


x: $\int \frac{(c+dx)^m}{a+b(F^g(e+fx))^n} dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\frac{1}{a+b(F^g(e+fx))^n} = -\partial_x \frac{\text{Log}\left[1 + \frac{a}{b(F^g(e+fx))^n}\right]}{a f g n \text{Log}[F]}$

Rule: If $m \in \mathbb{Z}^+$, then

$$\int \frac{(c+dx)^m}{a+b(F^g(e+fx))^n} dx \rightarrow -\frac{(c+dx)^m}{a f g n \text{Log}[F]} \text{Log}\left[1 + \frac{a}{b(F^g(e+fx))^n}\right] + \frac{d m}{a f g n \text{Log}[F]} \int (c+dx)^{m-1} \text{Log}\left[1 + \frac{a}{b(F^g(e+fx))^n}\right] dx$$

Program code:

```
(* Int[(c_+d_*x_)^m_/(a_+b_*(F^(g_*(e_+f_*x_)))^n_),x_Symbol] :=
  -(c+d*x)^m/(a*f*g*n*Log[F])*Log[1+a/(b*(F^(g*(e+f*x)))^n)] +
  d*m/(a*f*g*n*Log[F])*Int[(c+d*x)^(m-1)*Log[1+a/(b*(F^(g*(e+f*x)))^n)],x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && IGtQ[m,0] *)
```

4: $\int (c+dx)^m (a+b(F^g(e+fx))^n)^p dx$ when $p \in \mathbb{Z}^- \wedge m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $(a+bz)^p = \frac{(a+bz)^{p+1}}{b} - \frac{az(a+bz)^p}{b}$

Rule: If $p \in \mathbb{Z}^- \wedge m \in \mathbb{Z}^+$, then

$$\int (c+dx)^m (a+b(F^g(e+fx))^n)^p dx \rightarrow \frac{1}{a} \int (c+dx)^m (a+b(F^g(e+fx))^n)^{p+1} dx - \frac{b}{a} \int (c+dx)^m (F^g(e+fx))^n (a+b(F^g(e+fx))^n)^p dx$$

Program code:

```
Int[(c_.+d_.**x_)^m_.*(a_+b_.*(F^(g_.*(e_.+f_.**x_)))^n_.)^p_,x_Symbol] :=
  1/a*Int[(c+d*x)^m*(a+b*(F^(g*(e+f*x)))^n)^(p+1),x] -
  b/a*Int[(c+d*x)^m*(F^(g*(e+f*x)))^n*(a+b*(F^(g*(e+f*x)))^n)^p,x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && ILtQ[p,0] && IGtQ[m,0]
```

5: $\int (c+dx)^m (a+b(F^g(e+fx))^n)^p dx$ when $m \in \mathbb{Z}^+ \wedge p < -1$

Derivation: Integration by parts

Rule: If $m \in \mathbb{Z}^+ \wedge p < -1$, let $u = \int (a+b(F^g(e+fx))^n)^p dx$, then

$$\int (c+dx)^m (a+b(F^g(e+fx))^n)^p dx \rightarrow u (c+dx)^m - dm \int u (c+dx)^{m-1} dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*(F^(g_.*(e_.+f_.*x_)))^n_.)^p_,x_Symbol] :=
  With[{u=IntHide[(a+b*(F^(g*(e+f*x)))^n)^p,x]},
    Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x] /;
    FreeQ[{F,a,b,c,d,e,f,g,n},x] && IGtQ[m,0] && LtQ[p,-1]
```

6. $\int u^m (a+b(F^g v)^n)^p dx$ when $v = e+fx \wedge u = (c+dx)^q$

1: $\int u^m (a+b(F^g v)^n)^p dx$ when $v = e+fx \wedge u = (c+dx)^q \wedge m \in \mathbb{Z}$

Derivation: Algebraic normalization

Rule: If $v = e+fx \wedge u = (c+dx)^q \wedge m \in \mathbb{Z}$, then

$$\int u^m (a+b(F^g v)^n)^p dx \rightarrow \int (c+dx)^{mq} (a+b(F^g(e+fx))^n)^p dx$$

Program code:

```
Int[u_^m_.*(a_.+b_.*(F^(g_.*v_)) ^n_.)^p_,x_Symbol] :=
  Int[NormalizePowerOfLinear[u,x]^m*(a+b*(F^(g*ExpandToSum[v,x]))^n)^p,x] /;
  FreeQ[{F,a,b,g,n,p},x] && LinearQ[v,x] && PowerOfLinearQ[u,x] && Not[LinearMatchQ[v,x] && PowerOfLinearMatchQ[u,x]] && IntegerQ[m]
```

2: $\int u^m (a + b (F^g v)^n)^p dx$ when $v = e + f x \wedge u = (c + d x)^q \wedge m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{((c+dx)^q)^m}{(c+dx)^{mq}} = 0$

Rule: If $v = e + f x \wedge u = (c + d x)^q \wedge m \notin \mathbb{Z}$, then

$$\int u^m (a + b (F^g v)^n)^p dx \rightarrow \frac{((c + d x)^q)^m}{(c + d x)^{mq}} \int (c + d x)^{mq} (a + b (F^g (e + f x))^n)^p dx$$

Program code:

```
Int[u^m_.*(a_+b_.*(F^(g_.*v_))^n_.)^p_,x_Symbol] :=
Module[{uu=NormalizePowerOfLinear[u,x],z},
z=If[PowerQ[uu] && FreeQ[uu[[2]],x], uu[[1]]^(m*uu[[2]]), uu^m];
uu^m/z*Int[z*(a+b*(F^(g*ExpandToSum[v,x]))^n)^p,x] /;
FreeQ[{F,a,b,g,m,n,p},x] && LinearQ[v,x] && PowerOfLinearQ[u,x] && Not[LinearMatchQ[v,x] && PowerOfLinearMatchQ[u,x]] &&
Not[IntegerQ[m]]
```

x: $\int (c + d x)^m (a + b (F^g(e + f x))^n)^p dx$

Rule:

$$\int (c + d x)^m (a + b (F^g(e + f x))^n)^p dx \rightarrow \int (c + d x)^m (a + b (F^g(e + f x))^n)^p dx$$

Program code:

```
Int[(c_.+d_.**x_)^m_.*(a_.+b_.*(F^(g_.*(e_.+f_.**x_)))^n_.)^p_,x_Symbol] :=
  Unintegrable[(c+d*x)^m*(a+b*(F^(g*(e+f*x)))^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x]
```