# Rules for integrating miscellaneous algebraic functions

1. $\int \frac{u}{e\sqrt{a+bx}+f\sqrt{c+dx}}\,dx$

   **1:** $\int \frac{u}{e\sqrt{a+bx}+f\sqrt{c+dx}}\,dx$ when $b c - a d \neq 0 \land a e^2 - c f^2 == 0$

Derivation: Algebraic expansion

Basis: If $a e^2 - c f^2 == 0$, then $\frac{1}{e\sqrt{a+bx}+f\sqrt{c+dx}} == \frac{c\sqrt{a+bx}}{e(bc-ad)x} - \frac{a\sqrt{c+dx}}{f(bc-ad)x}$

Rule 1.3.3.1.1: If $b c - a d \neq 0 \land a e^2 - c f^2 == 0$, then

$$\int \frac{u}{e\sqrt{a+bx}+f\sqrt{c+dx}}\,dx \to \frac{c}{e(bc-ad)}\int \frac{u\sqrt{a+bx}}{x}\,dx - \frac{a}{f(bc-ad)}\int \frac{u\sqrt{c+dx}}{x}\,dx$$

Program code:

```
Int[u_/(e_.*Sqrt[a_.+b_.*x_]+f_.*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  c/(e*(b*c-a*d))*Int[(u*Sqrt[a+b*x])/x,x] - a/(f*(b*c-a*d))*Int[(u*Sqrt[c+d*x])/x,x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a*e^2-c*f^2,0]
```

**2:** $\int \frac{u}{e\sqrt{a+bx}+f\sqrt{c+dx}}\,dx$ when $bc-ad\neq 0\land be^2-df^2==0$

## Derivation: Algebraic expansion

Basis: If $be^2-df^2==0$, then $\frac{1}{e\sqrt{a+bx}+f\sqrt{c+dx}}==-\frac{d\sqrt{a+bx}}{e(bc-ad)}+\frac{b\sqrt{c+dx}}{f(bc-ad)}$

Rule 1.3.3.1.2: If $bc-ad\neq 0\land be^2-df^2==0$, then

$$\int \frac{u}{e\sqrt{a+bx}+f\sqrt{c+dx}}\,dx\longrightarrow -\frac{d}{e(bc-ad)}\int u\sqrt{a+bx}\,dx+\frac{b}{f(bc-ad)}\int u\sqrt{c+dx}\,dx$$

## Program code:

```
Int[u_/(e_.*Sqrt[a_.+b_.*x_]+f_.*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  -d/(e*(b*c-a*d))*Int[u*Sqrt[a+b*x],x] + b/(f*(b*c-a*d))*Int[u*Sqrt[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[b*e^2-d*f^2,0]
```

**3:** $\int \frac{u}{e \sqrt{a+b x}+f \sqrt{c+d x}} \, dx$ when $a e^2-c f^2 \neq 0 \land b e^2-d f^2 \neq 0$

## Derivation: Algebraic expansion

Basis: $\frac{1}{e \sqrt{a+b x}+f \sqrt{c+d x}} = \frac{e \sqrt{a+b x}}{a e^2-c f^2+\left(b e^2-d f^2\right) x}-\frac{f \sqrt{c+d x}}{a e^2-c f^2+\left(b e^2-d f^2\right) x}$

## Rule 1.3.3.1.3: If $a e^2-c f^2 \neq 0 \land b e^2-d f^2 \neq 0$, then

$$\int \frac{u}{e \sqrt{a+b x}+f \sqrt{c+d x}} \, dx \rightarrow e \int \frac{u \sqrt{a+b x}}{a e^2-c f^2+\left(b e^2-d f^2\right) x} \, dx-f \int \frac{u \sqrt{c+d x}}{a e^2-c f^2+\left(b e^2-d f^2\right) x} \, dx$$

## Program code:

```
Int[u_/(e_.*Sqrt[a_.+b_.*x_]+f_.*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  e*Int[(u*Sqrt[a+b*x])/(a*e^2-c*f^2+(b*e^2-d*f^2)*x),x] -
  f*Int[(u*Sqrt[c+d*x])/(a*e^2-c*f^2+(b*e^2-d*f^2)*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[a*e^2-c*f^2,0] && NeQ[b*e^2-d*f^2,0]
```

**2.** $\int \frac{u}{d\,x^n + c\,\sqrt{a + b\,x^{2\,n}}}\,dx$

**1:** $\int \frac{u}{d\,x^n + c\,\sqrt{a + b\,x^{2\,n}}}\,dx$ when $b\,c^2 - d^2 == 0$

Derivation: Algebraic expansion

Basis: If $b\,c^2 - d^2 == 0$, then $\frac{1}{d\,x^n + c\,\sqrt{a + b\,x^{2\,n}}} == -\frac{b\,x^n}{a\,d} + \frac{\sqrt{a + b\,x^{2\,n}}}{a\,c}$

Rule 1.3.3.2.1: If $b\,c^2 - d^2 == 0$, then

$$\int \frac{u}{d\,x^n + c\,\sqrt{a + b\,x^{2\,n}}}\,dx \longrightarrow -\frac{b}{a\,d}\int u\,x^n\,dx + \frac{1}{a\,c}\int u\,\sqrt{a + b\,x^{2\,n}}\,dx$$

Program code:

```
Int[u_./(d_.*x_^n_.+c_.*Sqrt[a_.+b_.*x_^p_.]),x_Symbol] :=
  -b/(a*d)*Int[u*x^n,x] + 1/(a*c)*Int[u*Sqrt[a+b*x^(2*n)],x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[p,2*n] && EqQ[b*c^2-d^2,0]
```

**2:** $\int \frac{x^m}{d\,x^n + c\,\sqrt{a + b\,x^{2\,n}}}\,dx$ when $b\,c^2 - d^2 \neq 0$

## Derivation: Algebraic expansion

Basis: $\frac{1}{d\,x^n + c\,\sqrt{a+b\,x^{2\,n}}} = -\frac{d\,x^n}{a\,c^2 + (b\,c^2 - d^2)\,x^{2\,n}} + \frac{c\,\sqrt{a+b\,x^{2\,n}}}{a\,c^2 + (b\,c^2 - d^2)\,x^{2\,n}}$

Rule 1.3.3.2.2: If $b\,c^2 - d^2 \neq 0$, then

$$\int \frac{x^m}{d\,x^n + c\,\sqrt{a + b\,x^{2\,n}}}\,dx \longrightarrow -d\int \frac{x^{m+n}}{a\,c^2 + (b\,c^2 - d^2)\,x^{2\,n}}\,dx + c\int \frac{x^m\,\sqrt{a + b\,x^{2\,n}}}{a\,c^2 + (b\,c^2 - d^2)\,x^{2\,n}}\,dx$$

## Program code:

```
Int[x_^m_./(d_.*x_^n_.+c_.*Sqrt[a_.+b_.*x_^p_.]),x_Symbol] :=
  -d*Int[x^(m+n)/(a*c^2+(b*c^2-d^2)*x^(2*n)),x] +
  c*Int[(x^m*Sqrt[a+b*x^(2*n)])/(a*c^2+(b*c^2-d^2)*x^(2*n)),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[p,2*n] && NeQ[b*c^2-d^2,0]
```

**3.** $\int \frac{1}{(a + b\,x^3)\,\sqrt{d + e\,x + f\,x^2}}\,dx$

**1:** $\int \frac{1}{(a + b\,x^3)\,\sqrt{d + e\,x + f\,x^2}}\,dx$ when $\frac{a}{b} > 0$

## Derivation: Algebraic expansion

Basis: If $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/3}$, then $\frac{1}{a+b\,z^3} = \frac{r}{3\,a\,(r+s\,z)} + \frac{r\,(2\,r-s\,z)}{3\,a\,(r^2-r\,s\,z+s^2\,z^2)}$

Rule 1.3.3.3.1: If $\frac{a}{b} > 0$, let $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/3}$, then

$$\int \frac{1}{\left(a+b x^3\right) \sqrt{d+e x+f x^2}} \, dx \to \frac{r}{3 a} \int \frac{1}{(r+s x) \sqrt{d+e x+f x^2}} \, dx+\frac{r}{3 a} \int \frac{2 r-s x}{\left(r^2-r s x+s^2 x^2\right) \sqrt{d+e x+f x^2}} \, dx$$

## Program code:

```
Int[1/((a_+b_.*x_^3)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,3]], s=Denominator[Rt[a/b,3]]},
  r/(3*a)*Int[1/((r+s*x)*Sqrt[d+e*x+f*x^2]),x] +
  r/(3*a)*Int[(2*r-s*x)/((r^2-r*s*x+s^2*x^2)*Sqrt[d+e*x+f*x^2]),x]] /;
FreeQ[{a,b,d,e,f},x] && PosQ[a/b]
```

```
Int[1/((a_+b_.*x_^3)*Sqrt[d_.+f_.*x_^2]),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,3]], s=Denominator[Rt[a/b,3]]},
  r/(3*a)*Int[1/((r+s*x)*Sqrt[d+f*x^2]),x] +
  r/(3*a)*Int[(2*r-s*x)/((r^2-r*s*x+s^2*x^2)*Sqrt[d+f*x^2]),x]] /;
FreeQ[{a,b,d,f},x] && PosQ[a/b]
```

**2:** $\int \frac{1}{\left(a+b x^3\right) \sqrt{d+e x+f x^2}} \, dx$ when $\frac{a}{b} \ngtr 0$

## Derivation: Algebraic expansion

Basis: If $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$, then $\frac{1}{a+b z^3} = \frac{r}{3 a (r-s z)} + \frac{r (2 r+s z)}{3 a \left(r^2+r s z+s^2 z^2\right)}$

Rule 1.3.3.3.2: If $\frac{a}{b} \ngtr 0$, let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$, then

$$\int \frac{1}{\left(a+b x^3\right) \sqrt{d+e x+f x^2}} \, dx \to \frac{r}{3 a} \int \frac{1}{(r-s x) \sqrt{d+e x+f x^2}} \, dx + \frac{r}{3 a} \int \frac{2 r+s x}{\left(r^2+r s x+s^2 x^2\right) \sqrt{d+e x+f x^2}} \, dx$$

## Program code:

```
Int[1/((a_+b_.*x_^3)*Sqrt[d_.+e_.*x_+f_.*x_^2]),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,3]], s=Denominator[Rt[-a/b,3]]},
  r/(3*a)*Int[1/((r-s*x)*Sqrt[d+e*x+f*x^2]),x] +
  r/(3*a)*Int[(2*r+s*x)/((r^2+r*s*x+s^2*x^2)*Sqrt[d+e*x+f*x^2]),x]] /;
FreeQ[{a,b,d,e,f},x] && NegQ[a/b]
```

```
Int[1/((a_+b_.*x_^3)*Sqrt[d_.+f_.*x_^2]),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,3]], s=Denominator[Rt[-a/b,3]]},
  r/(3*a)*Int[1/((r-s*x)*Sqrt[d+f*x^2]),x] +
  r/(3*a)*Int[(2*r+s*x)/((r^2+r*s*x+s^2*x^2)*Sqrt[d+f*x^2]),x]] /;
FreeQ[{a,b,d,f},x] && NegQ[a/b]
```

**4:** $\displaystyle\int \frac{A+B x^4}{\left(d+e x^2+f x^4\right) \sqrt{a+b x^2+c x^4}}\,dx$ when $a B+A c==0 \wedge c d-a f==0$

## Derivation: Integration by substitution

Basis: If $a B+A c==0 \wedge c d-a f==0$, then $\frac{A+B x^4}{\left(d+e x^2+f x^4\right) \sqrt{a+b x^2+c x^4}}==A\, \text{Subst}\left[\frac{1}{d-(b d-a e)\, x^2},\, x,\, \frac{x}{\sqrt{a+b x^2+c x^4}}\right] \partial_x \frac{x}{\sqrt{a+b x^2+c x^4}}$

Rule 1.3.3.4: If $a B+A c==0 \wedge c d-a f==0$, then

$$\int \frac{A+B x^4}{\left(d+e x^2+f x^4\right) \sqrt{a+b x^2+c x^4}}\,dx \rightarrow A\, \text{Subst}\left[\int \frac{1}{d-(b d-a e)\, x^2}\,dx,\, x,\, \frac{x}{\sqrt{a+b x^2+c x^4}}\right]$$

## Program code:

```
Int[u_*(A_+B_.*x_^4)/Sqrt[v_],x_Symbol] :=
  With[{a=Coeff[v,x,0],b=Coeff[v,x,2],c=Coeff[v,x,4],d=Coeff[1/u,x,0],e=Coeff[1/u,x,2],f=Coeff[1/u,x,4]},
  A*Subst[Int[1/(d-(b*d-a*e)*x^2),x],x,x/Sqrt[v]] /;
 EqQ[a*B+A*c,0] && EqQ[c*d-a*f,0]] /;
FreeQ[{A,B},x] && PolyQ[v,x^2,2] && PolyQ[1/u,x^2,2]
```

**5:** $\int \frac{1}{(a+b\,x)\,\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\,x} = \frac{a}{a^2-b^2\,x^2} - \frac{b\,x}{a^2-b^2\,x^2}$

Rule 1.3.3.5:

$$\int \frac{1}{(a+b\,x)\,\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}\,dx \to a\int \frac{1}{(a^2-b^2\,x^2)\,\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}\,dx - b\int \frac{x}{(a^2-b^2\,x^2)\,\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}\,dx$$

Program code:

```
Int[1/((a_+b_.*x_)*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  a*Int[1/((a^2-b^2*x^2)*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] - b*Int[x/((a^2-b^2*x^2)*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x]
```

**6.** $\int u\left(d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right)^n dx$ when $d^2-a\,f^2=0$

**1:** $\int (g+h\,x)\,\sqrt{d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $(e\,g-d\,h)^2-f^2\,(c\,g^2-b\,g\,h+a\,h^2)=0\,\land\,2\,e^2\,g-2\,d\,e\,h-f^2\,(2\,c\,g-b\,h)=0$

Author: Martin Welz via email on 21 July 2014

Derivation: Integration by substitution

Rule 1.3.3.6.1: If $(e\,g-d\,h)^2-f^2\,(c\,g^2-b\,g\,h+a\,h^2)=0\,\land\,2\,e^2\,g-2\,d\,e\,h-f^2\,(2\,c\,g-b\,h)=0$, then

$$\int (g+h\,x)\,\sqrt{d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}}\,dx \to$$

$$\frac{1}{15\,c^2\,f\,(g+h\,x)}2\left(f\,\left(5\,b\,c\,g^2-2\,b^2\,g\,h-3\,a\,c\,g\,h+2\,a\,b\,h^2\right)+\right.$$
$$\left.c\,f\,\left(10\,c\,g^2-b\,g\,h+a\,h^2\right)\,x+9\,c^2\,f\,g\,h\,x^2+3\,c^2\,f\,h^2\,x^3-(e\,g-d\,h)\,(5\,c\,g-2\,b\,h+c\,h\,x)\,\sqrt{a+b\,x+c\,x^2}\right)\sqrt{d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}}$$

## Program code:

```
Int[(g_.+h_.*x_)*Sqrt[d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2]],x_Symbol] :=
  2*(f*(5*b*c*g^2-2*b^2*g*h-3*a*c*g*h+2*a*b*h^2)+c*f*(10*c*g^2-b*g*h+a*h^2)*x+9*c^2*f*g*h*x^2+3*c^2*f*h^2*x^3-
    (e*g-d*h)*(5*c*g-2*b*h+c*h*x)*Sqrt[a+b*x+c*x^2])/
   (15*c^2*f*(g+h*x))*Sqrt[d+e*x+f*Sqrt[a+b*x+c*x^2]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && EqQ[(e*g-d*h)^2-f^2*(c*g^2-b*g*h+a*h^2),0] && EqQ[2*e^2*g-2*d*e*h-f^2*(2*c*g-b*h),0]
```

**2:** $\int (g+h\,x)^m\,\left(u+f\,\left(j+k\,\sqrt{v}\right)\right)^n\,dx$ when $u == d+e\,x\ \land\ v == a+b\,x+c\,x^2\ \land\ (e\,g-h\,(d+f\,j))^2-f^2\,k^2\,\left(c\,g^2-b\,g\,h+a\,h^2\right) == 0$

## Derivation: Algebraic normalization

## Rule 1.3.3.6.2: If $u == d+e\,x\ \land\ v == a+b\,x+c\,x^2\ \land\ (e\,g-h\,(d+f\,j))^2-f^2\,k^2\,\left(c\,g^2-b\,g\,h+a\,h^2\right) == 0$, then

$$\int (g+h\,x)^m\,\left(u+f\,\left(j+k\,\sqrt{v}\right)\right)^n\,dx\ \longrightarrow\ \int (g+h\,x)^m\,\left(d+f\,j+e\,x+f\,k\,\sqrt{a+b\,x+c\,x^2}\right)^n\,dx$$

## Program code:

```
Int[(g_.+h_.*x_)^m_.*(u_+f_.*(j_.+k_.*Sqrt[v_]))^n_.,x_Symbol] :=
  Int[(g+h*x)^m*(ExpandToSum[u+f*j,x]+f*k*Sqrt[ExpandToSum[v,x]])^n,x] /;
FreeQ[{f,g,h,j,k,m,n},x] && LinearQ[u,x] && QuadraticQ[v,x] &&
  Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x] && (EqQ[j,0] || EqQ[f,1])] &&
  EqQ[(Coefficient[u,x,1]*g-h*(Coefficient[u,x,0]+f*j))^2-f^2*k^2*(Coefficient[v,x,2]*g^2-Coefficient[v,x,1]*g*h+Coefficient[v,x,0]*h^
```

**7.** $\int u\,\left(d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right)^n\,dx$ when $e^2-c\,f^2 == 0$

**x:** $\int \frac{1}{d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $e^2-c\,f^2==0$

Derivation: Algebraic expansion

Basis: If $e^2-c\,f^2==0$, then $\frac{1}{d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}}==\frac{d+e\,x-f\,\sqrt{a+b\,x+c\,x^2}}{d^2-a\,f^2+\left(2\,d\,e-b\,f^2\right)\,x}==\frac{d+e\,x}{d^2-a\,f^2+\left(2\,d\,e-b\,f^2\right)\,x}-\frac{f\,\sqrt{a+b\,x+c\,x^2}}{d^2-a\,f^2+\left(2\,d\,e-b\,f^2\right)\,x}$

Note: Unfortunately this does not give as simple an antiderivative as the Euler substitution.

Rule 1.3.3.7.x: If $e^2-c\,f^2==0$, then

$$\int \frac{1}{d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}}\,dx\to\int \frac{d+e\,x}{d^2-a\,f^2+\left(2\,d\,e-b\,f^2\right)\,x}\,dx-f\int \frac{\sqrt{a+b\,x+c\,x^2}}{d^2-a\,f^2+\left(2\,d\,e-b\,f^2\right)\,x}\,dx$$

Program code:

```
(* Int[1/(d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  Int[(d+e*x)/(d^2-a*f^2+(2*d*e-b*f^2)*x),x] -
  f*Int[Sqrt[a+b*x+c*x^2]/(d^2-a*f^2+(2*d*e-b*f^2)*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e^2-c*f^2,0] *)
```

```
(* Int[1/(d_.+e_.*x_+f_.*Sqrt[a_.+c_.*x_^2]),x_Symbol] :=
  Int[(d+e*x)/(d^2-a*f^2+2*d*e*x),x] -
  f*Int[Sqrt[a+c*x^2]/(d^2-a*f^2+2*d*e*x),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[e^2-c*f^2,0] *)
```

1. $\int \left(g+h\left(d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right)^n\right)^p\,dx$ when $e^2-c\,f^2==0$

**1:** $\int \left(g+h\left(d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right)^n\right)^p\,dx$ when $e^2-c\,f^2==0\,\land\,p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $e^2-c\,f^2==0$, then

$$1 == 2\,\text{Subst}\left[\frac{\left(d^2 e-(b d-a e) f^2-\left(2 d e-b f^2\right) x+e x^2\right)}{\left(-2 d e+b f^2+2 e x\right)^2},\,x,\,d+e x+f \sqrt{a+b x+c x^2}\right] \partial_x \left(d+e x+f \sqrt{a+b x+c x^2}\right)$$

Note: This is a special case of Euler substitution #2

Rule 1.3.3.7.1.1: If $e^2-c f^2==0 \land p\in\mathbb{Z}$, then

$$\int\left(g+h\left(d+e x+f \sqrt{a+b x+c x^2}\right)^n\right)^p dx \to 2\,\text{Subst}\left[\int\frac{\left(g+h x^n\right)^p \left(d^2 e-(b d-a e) f^2-\left(2 d e-b f^2\right) x+e x^2\right)}{\left(-2 d e+b f^2+2 e x\right)^2} dx,\,x,\,d+e x+f \sqrt{a+b x+c x^2}\right]$$

Program code:

```
Int[(g_.+h_.*(d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2])^n_)^p_.,x_Symbol] :=
  2*Subst[Int[(g+h*x^n)^p*(d^2*e-(b*d-a*e)*f^2-(2*d*e-b*f^2)*x+e*x^2)/(-2*d*e+b*f^2+2*e*x)^2,x],x,d+e*x+f*Sqrt[a+b*x+c*x^2]] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x] && EqQ[e^2-c*f^2,0] && IntegerQ[p]
```

```
Int[(g_.+h_.*(d_.+e_.*x_+f_.*Sqrt[a_+c_.*x_^2])^n_)^p_.,x_Symbol] :=
  1/(2*e)*Subst[Int[(g+h*x^n)^p*(d^2+a*f^2-2*d*x+x^2)/(d-x)^2,x],x,d+e*x+f*Sqrt[a+c*x^2]] /;
FreeQ[{a,c,d,e,f,g,h,n},x] && EqQ[e^2-c*f^2,0] && IntegerQ[p]
```

**2:** $\int \left(g+h\left(u+f\sqrt{v}\right)^n\right)^p dx$ when $u == d+e\,x \land v == a+b\,x+c\,x^2 \land e^2-c\,f^2 == 0 \land p \in \mathbb{Z}$

## Derivation: Algebraic normalization

## Rule 1.3.3.7.1.2: If $u == d+e\,x \land v == a+b\,x+c\,x^2 \land e^2-c\,f^2 == 0 \land p \in \mathbb{Z}$, then

$$\int \left(g+h\left(u+f\sqrt{v}\right)^n\right)^p dx \to \int \left(g+h\left(d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right)^n\right)^p dx$$

## Program code:

```
Int[(g_.+h_.*(u_+f_.Sqrt[v_])^n_)^p_.,x_Symbol] :=
  Int[(g+h*(ExpandToSum[u,x]+f*Sqrt[ExpandToSum[v,x]])^n)^p,x] /;
FreeQ[{f,g,h,n},x] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]] &&
  EqQ[Coefficient[u,x,1]^2-Coefficient[v,x,2]*f^2,0] && IntegerQ[p]
```

**2:** $\int (g+h\,x)^m \left(e\,x+f\sqrt{a+c\,x^2}\right)^n dx$ when $e^2-c\,f^2 == 0 \land m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $e^2-c\,f^2 == 0 \land m \in \mathbb{Z}$, then

$$(g+h\,x)^m == \frac{1}{2^{m+1}\,e^{m+1}}\,\text{Subst}\left[\frac{\left(a\,f^2+x^2\right)\left(-a\,f^2\,h+2\,e\,g\,x+h\,x^2\right)^m}{x^{m+2}},\,x,\,e\,x+f\sqrt{a+c\,x^2}\right]\partial_x\left(e\,x+f\sqrt{a+c\,x^2}\right)$$

Note: This is a special case of Euler substitution #2

Rule 1.3.3.7.2: If $e^2-c\,f^2 == 0 \land m \in \mathbb{Z}$, then

$$\int (g+h\,x)^m \left(e\,x+f\sqrt{a+c\,x^2}\right)^n dx \longrightarrow \frac{1}{2^{m+1}\,e^{m+1}}\,\text{Subst}\left[\int x^{n-m-2}\left(a\,f^2+x^2\right)\left(-a\,f^2\,h+2\,e\,g\,x+h\,x^2\right)^m dx,\,x,\,e\,x+f\sqrt{a+c\,x^2}\right]$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*(e_.*x_+f_.*Sqrt[a_.+c_.*x_^2])^n_.,x_Symbol] :=
  1/(2^(m+1)*e^(m+1))*Subst[Int[x^(n-m-2)*(a*f^2+x^2)*(-a*f^2*h+2*e*g*x+h*x^2)^m,x],x,e*x+f*Sqrt[a+c*x^2]] /;
FreeQ[{a,c,e,f,g,h,n},x] && EqQ[e^2-c*f^2,0] && IntegerQ[m]
```

3: $\int x^p \left(g+i x^2\right)^m \left(e x+f \sqrt{a+c x^2}\right)^n dx$ when $e^2-c f^2=0 \land c g-a i=0 \land (p \mid 2 m) \in \mathbb{Z} \land \left(m \in \mathbb{Z} \lor \frac{i}{c}>0\right)$

Derivation: Integration by substitution

Basis: If $e^2-c f^2=0 \land c g-a i=0 \land (p \mid 2 m) \in \mathbb{Z} \land \left(m \in \mathbb{Z} \lor \frac{i}{c}>0\right)$, then

$$x^p \left(g+i x^2\right)^m = \left(\frac{i}{c}\right)^m x^p \left(a+c x^2\right)^m =$$

$$\frac{1}{2^{2 m+p+1} e^{p+1} f^{2 m}} \left(\frac{i}{c}\right)^m \text{Subst}\left[\frac{\left(-a f^2+x^2\right)^p \left(a f^2+x^2\right)^{2 m+1}}{x^{2 m+p+2}},\ x,\ e x+f \sqrt{a+c x^2}\right] \partial_x \left(e x+f \sqrt{a+c x^2}\right)$$

Note: This is a special case of Euler substitution #2

Rule 1.3.3.7.3: If $e^2-c f^2=0 \land c g-a i=0 \land (p \mid 2 m) \in \mathbb{Z} \land \left(m \in \mathbb{Z} \lor \frac{i}{c}>0\right)$, then

$$\int x^p \left(g+i x^2\right)^m \left(e x+f \sqrt{a+c x^2}\right)^n dx \to \frac{1}{2^{2 m+p+1} e^{p+1} f^{2 m}} \left(\frac{i}{c}\right)^m \text{Subst}\left[\int x^{n-2 m-p-2} \left(-a f^2+x^2\right)^p \left(a f^2+x^2\right)^{2 m+1} dx,\ x,\ e x+f \sqrt{a+c x^2}\right]$$

Program code:

```
Int[x_^p_.*(g_+i_.*x_^2)^m_.*(e_.*x_+f_.*Sqrt[a_+c_.*x_^2])^n_.,x_Symbol] :=
  1/(2^(2*m+p+1)*e^(p+1)*f^(2*m))*(i/c)^m*Subst[Int[x^(n-2*m-p-2)*(-a*f^2+x^2)^p*(a*f^2+x^2)^(2*m+1),x],x,e*x+f*Sqrt[a+c*x^2]] /;
FreeQ[{a,c,e,f,g,i,n},x] && EqQ[e^2-c*f^2,0] && EqQ[c*g-a*i,0] && IntegersQ[p,2*m] && (IntegerQ[m] || GtQ[i/c,0])
```

4. $\int \left(g+h x+i x^2\right)^m \left(d+e x+f \sqrt{a+b x+c x^2}\right)^n dx$ when $e^2-c f^2=0$

1: $\int \left(g+h x+i x^2\right)^m \left(d+e x+f \sqrt{a+b x+c x^2}\right)^n dx$ when $e^2-c f^2=0 \land c g-a i=0 \land c h-b i=0 \land 2 m \in \mathbb{Z} \land \left(m \in \mathbb{Z} \lor \frac{i}{c}>0\right)$

Derivation: Integration by substitution

Basis: If $e^2-c f^2=0 \land c g-a i=0 \land c h-b i=0 \land 2 m \in \mathbb{Z} \land \left(m \in \mathbb{Z} \lor \frac{i}{c}>0\right)$, then

$$\left(g+h\,x+i\,x^2\right)^m == \left(\frac{i}{c}\right)^m \left(a+b\,x+c\,x^2\right)^m ==$$

$$\frac{2}{f^{2\,m}}\,\left(\frac{i}{c}\right)^m \text{Subst}\left[\frac{\left(d^2\,e-(b\,d-a\,e)\,f^2-\left(2\,d\,e-b\,f^2\right)\,x+e\,x^2\right)^{2\,m+1}}{\left(-2\,d\,e+b\,f^2+2\,e\,x\right)^{2\,(m+1)}},\,x,\,d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right]$$

$$\partial_x\left(d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right)$$

Note: This is a special case of Euler substitution #2

Rule 1.3.3.7.4.1: If $e^2-c\,f^2==0\,\wedge\,c\,g-a\,i==0\,\wedge\,c\,h-b\,i==0\,\wedge\,2\,m\in\mathbb{Z}\,\wedge\,\left(m\in\mathbb{Z}\,\vee\,\frac{i}{c}>0\right)$, then

$$\int\left(g+h\,x+i\,x^2\right)^m\left(d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right)^n dx \,\rightarrow\, \frac{2}{f^{2\,m}}\left(\frac{i}{c}\right)^m \text{Subst}\left[\int\frac{x^n\left(d^2\,e-(b\,d-a\,e)\,f^2-\left(2\,d\,e-b\,f^2\right)\,x+e\,x^2\right)^{2\,m+1}}{\left(-2\,d\,e+b\,f^2+2\,e\,x\right)^{2\,(m+1)}}\,dx,\,x,\,d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\right]$$

Program code:

```
Int[(g_.+h_.*x_+i_.*x_^2)^m_.*(d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2])^n_.,x_Symbol] :=
  2/f^(2*m)*(i/c)^m*
    Subst[Int[x^n*(d^2*e-(b*d-a*e)*f^2-(2*d*e-b*f^2)*x+e*x^2)^(2*m+1)/(-2*d*e+b*f^2+2*e*x)^(2*(m+1)),x],x,d+e*x+f*Sqrt[a+b*x+c*x^2]] /
FreeQ[{a,b,c,d,e,f,g,h,i,n},x] && EqQ[e^2-c*f^2,0] && EqQ[c*g-a*i,0] && EqQ[c*h-b*i,0] && IntegerQ[2*m] && (IntegerQ[m] || GtQ[i/c,0])
```

```
Int[(g_+i_.*x_^2)^m_.*(d_.+e_.*x_+f_.*Sqrt[a_+c_.*x_^2])^n_.,x_Symbol] :=
  1/(2^(2*m+1)*e*f^(2*m))*(i/c)^m*
    Subst[Int[x^n*(d^2+a*f^2-2*d*x+x^2)^(2*m+1)/(-d+x)^(2*(m+1)),x],x,d+e*x+f*Sqrt[a+c*x^2]] /;
FreeQ[{a,c,d,e,f,g,i,n},x] && EqQ[e^2-c*f^2,0] && EqQ[c*g-a*i,0] && IntegerQ[2*m] && (IntegerQ[m] || GtQ[i/c,0])
```

2. $\int (g + h x + i x^2)^m \left(d + e x + f \sqrt{a + b x + c x^2}\right)^n dx$ when $e^2 - c f^2 == 0 \land c g - a i == 0 \land c h - b i == 0 \land m + \frac{1}{2} \in \mathbb{Z} \land \frac{i}{c} \not> 0$

**1:** $\int (g + h x + i x^2)^m \left(d + e x + f \sqrt{a + b x + c x^2}\right)^n dx$ when $e^2 - c f^2 == 0 \land c g - a i == 0 \land c h - b i == 0 \land m + \frac{1}{2} \in \mathbb{Z}^+ \land \frac{i}{c} \not> 0$

## Derivation: Piecewise constant extraction

## Basis: If $c g - a i == 0 \land c h - b i == 0$, then $\partial_x \frac{\sqrt{g+h x+i x^2}}{\sqrt{a+b x+c x^2}} == 0$

## Rule 1.3.3.7.4.2.1: If $e^2 - c f^2 == 0 \land c g - a i == 0 \land c h - b i == 0 \land m + \frac{1}{2} \in \mathbb{Z}^+ \land \frac{i}{c} \not> 0$, then

$$\int (g + h x + i x^2)^m \left(d + e x + f \sqrt{a + b x + c x^2}\right)^n dx \rightarrow \left(\frac{i}{c}\right)^{m-\frac{1}{2}} \frac{\sqrt{g + h x + i x^2}}{\sqrt{a + b x + c x^2}} \int (a + b x + c x^2)^m \left(d + e x + f \sqrt{a + b x + c x^2}\right)^n dx$$

## Program code:

```
Int[(g_.+h_.*x_+i_.*x_^2)^m_.*(d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2])^n_.,x_Symbol] :=
  (i/c)^(m-1/2)*Sqrt[g+h*x+i*x^2]/Sqrt[a+b*x+c*x^2]*Int[(a+b*x+c*x^2)^m*(d+e*x+f*Sqrt[a+b*x+c*x^2])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,n},x] && EqQ[e^2-c*f^2,0] && EqQ[c*g-a*i,0] && EqQ[c*h-b*i,0] && IGtQ[m+1/2,0] && Not[GtQ[i/c,0]]
```

```
Int[(g_+i_.*x_^2)^m_.*(d_.+e_.*x_+f_.*Sqrt[a_+c_.*x_^2])^n_.,x_Symbol] :=
  (i/c)^(m-1/2)*Sqrt[g+i*x^2]/Sqrt[a+c*x^2]*Int[(a+c*x^2)^m*(d+e*x+f*Sqrt[a+c*x^2])^n,x] /;
FreeQ[{a,c,d,e,f,g,i,n},x] && EqQ[e^2-c*f^2,0] && EqQ[c*g-a*i,0] && IGtQ[m+1/2,0] && Not[GtQ[i/c,0]]
```

**2:** $\int (g + h x + i x^2)^m \left(d + e x + f \sqrt{a + b x + c x^2}\right)^n dx$ when $e^2 - c f^2 == 0 \land c g - a i == 0 \land c h - b i == 0 \land m - \frac{1}{2} \in \mathbb{Z}^- \land \frac{i}{c} \ngtr 0$

## Derivation: Piecewise constant extraction

Basis: If $c g - a i == 0 \land c h - b i == 0$, then $\partial_x \frac{\sqrt{a+b x+c x^2}}{\sqrt{g+h x+i x^2}} == 0$

Rule 1.3.3.7.4.2.2: If $e^2 - c f^2 == 0 \land c g - a i == 0 \land c h - b i == 0 \land m - \frac{1}{2} \in \mathbb{Z}^- \land \frac{i}{c} \ngtr 0$, then

$$\int (g + h x + i x^2)^m \left(d + e x + f \sqrt{a + b x + c x^2}\right)^n dx \rightarrow \left(\frac{i}{c}\right)^{m+\frac{1}{2}} \frac{\sqrt{a + b x + c x^2}}{\sqrt{g + h x + i x^2}} \int (a + b x + c x^2)^m \left(d + e x + f \sqrt{a + b x + c x^2}\right)^n dx$$

## Program code:

```
Int[(g_.+h_.*x_+i_.*x_^2)^m_.*(d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2])^n_.,x_Symbol] :=
  (i/c)^(m+1/2)*Sqrt[a+b*x+c*x^2]/Sqrt[g+h*x+i*x^2]*Int[(a+b*x+c*x^2)^m*(d+e*x+f*Sqrt[a+b*x+c*x^2])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,n},x] && EqQ[e^2-c*f^2,0] && EqQ[c*g-a*i,0] && EqQ[c*h-b*i,0] && ILtQ[m-1/2,0] && Not[GtQ[i/c,0]]
```

```
Int[(g_+i_.*x_^2)^m_.*(d_.+e_.*x_+f_.*Sqrt[a_+c_.*x_^2])^n_.,x_Symbol] :=
  (i/c)^(m+1/2)*Sqrt[a+c*x^2]/Sqrt[g+i*x^2]*Int[(a+c*x^2)^m*(d+e*x+f*Sqrt[a+c*x^2])^n,x] /;
FreeQ[{a,c,d,e,f,g,i,n},x] && EqQ[e^2-c*f^2,0] && EqQ[c*g-a*i,0] && ILtQ[m-1/2,0] && Not[GtQ[i/c,0]]
```

3: $\int w^m \left(u+f\left(j+k\sqrt{v}\right)\right)^n dx$ when $u == d+e\,x \land v == a+b\,x+c\,x^2 \land w == g+h\,x+i\,x^2 \land e^2-c\,f^2\,k^2 == 0$

## Derivation: Algebraic normalization

## Rule 1.3.3.7.4.3: If $u == d+e\,x \land v == a+b\,x+c\,x^2 \land w == g+h\,x+i\,x^2 \land e^2-c\,f^2\,k^2 == 0$, then

$$\int w^m \left(u+f\left(j+k\sqrt{v}\right)\right)^n dx \longrightarrow \int \left(g+h\,x+i\,x^2\right)^m \left(d+f\,j+e\,x+f\,k\sqrt{a+b\,x+c\,x^2}\right)^n dx$$

## Program code:

```
Int[w_^m_.*(u_+f_.*(j_.+k_.*Sqrt[v_]))^n_.,x_Symbol] :=
  Int[ExpandToSum[w,x]^m*(ExpandToSum[u+f*j,x]+f*k*Sqrt[ExpandToSum[v,x]])^n,x] /;
FreeQ[{f,j,k,m,n},x] && LinearQ[u,x] && QuadraticQ[{v,w},x] &&
  Not[LinearMatchQ[u,x] && QuadraticMatchQ[{v,w},x] && (EqQ[j,0] || EqQ[f,1])] &&
  EqQ[Coefficient[u,x,1]^2-Coefficient[v,x,2]*f^2*k^2,0]
```

**8:** $\int \frac{1}{(a + b x^n) \sqrt{c x^2 + d (a + b x^n)^{2/n}}} \, dx$

Reference: Integration of Functions (1948) by A.F. Timofeev

Derivation: Integration by substitution

Basis: $\frac{1}{(a+b x^n) \sqrt{c x^2+d (a+b x^n)^{2/n}}} == \frac{1}{a} \text{Subst}\left[\frac{1}{1-c x^2}, x, \frac{x}{\sqrt{c x^2+d (a+b x^n)^{2/n}}}\right] \partial_x \frac{x}{\sqrt{c x^2+d (a+b x^n)^{2/n}}}$

Rule 1.3.3.8:

$$\int \frac{1}{(a + b x^n) \sqrt{c x^2 + d (a + b x^n)^{2/n}}} \, dx \to \frac{1}{a} \text{Subst}\left[\int \frac{1}{1 - c x^2} \, dx, x, \frac{x}{\sqrt{c x^2 + d (a + b x^n)^{2/n}}}\right]$$

Program code:

```
Int[1/((a_+b_.*x_^n_.)*Sqrt[c_.*x_^2+d_.*(a_+b_.*x_^n_.)^p_.]),x_Symbol] :=
  1/a*Subst[Int[1/(1-c*x^2),x],x,x/Sqrt[c*x^2+d*(a+b*x^n)^(2/n)]] /;
FreeQ[{a,b,c,d,n},x] && EqQ[p,2/n]
```

**9:** $\int \sqrt{a + b \sqrt{c + d x^2}} \, dx$ when $a^2 - b^2 c == 0$

Derivation: Integration by substitution

Basis: If $a^2 - b^2 c == 0$, then

$$\sqrt{a + b \sqrt{c + d x^2}} == -2 a \,\text{Subst}\left[\frac{b^2 d + x^2}{(b^2 d - x^2)^2} \sqrt{-\frac{2 a x^2}{b^2 d - x^2}}, x, \frac{a + b \sqrt{c + d x^2}}{x}\right] \partial_x \frac{a + b \sqrt{c + d x^2}}{x}$$

Note: This is a special case of Euler substitution #1, if $d^2 - f^2 a == 0$, then

$$\sqrt{d+f\sqrt{a+b\,x+c\,x^2}} == -2\,\text{Subst}\left[\frac{c\,d\,f^2+b\,f^2\,x+d\,x^2}{\left(c\,f^2-x^2\right)^2}\sqrt{d-\frac{c\,d\,f^2+b\,f^2\,x+d\,x^2}{c\,f^2-x^2}},\ x,\ \frac{d+f\sqrt{a+b\,x+c\,x^2}}{x}\right]\partial_x\frac{d+f\sqrt{a+b\,x+c\,x^2}}{x}$$

Rule 1.3.3.9: If $a^2-b^2\,c==0$, then

$$\int\sqrt{a+b\sqrt{c+d\,x^2}}\,dx\ \longrightarrow\ -2\,a\,\text{Subst}\left[\int\frac{b^2\,d+x^2}{\left(b^2\,d-x^2\right)^2}\sqrt{-\frac{2\,a\,x^2}{b^2\,d-x^2}}\,dx,\ x,\ \frac{a+b\sqrt{c+d\,x^2}}{x}\right]$$

$$\longrightarrow\ \frac{2\,b^2\,d\,x^3}{3\left(a+b\sqrt{c+d\,x^2}\right)^{3/2}}+\frac{2\,a\,x}{\sqrt{a+b\sqrt{c+d\,x^2}}}$$

Program code:

```
Int[Sqrt[a_+b_.*Sqrt[c_+d_.*x_^2]],x_Symbol] :=
  2*b^2*d*x^3/(3*(a+b*Sqrt[c+d*x^2])^(3/2)) + 2*a*x/Sqrt[a+b*Sqrt[c+d*x^2]] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2*c,0]
```

**10:** $\int\frac{\sqrt{a\,x^2+b\,x\sqrt{c+d\,x^2}}}{x\sqrt{c+d\,x^2}}\,dx$ when $a^2-b^2\,d==0\ \land\ b^2\,c+a==0$

Derivation: Integration by substitution

Basis: If $a^2-b^2\,d==0\ \land\ b^2\,c+a==0$, then

$$\frac{\sqrt{a\,x^2+b\,x\sqrt{c+d\,x^2}}}{x\sqrt{c+d\,x^2}} == \frac{\sqrt{2}\,b}{a}\,\text{Subst}\left[\frac{1}{\sqrt{1+\frac{x^2}{a}}},\ x,\ a\,x+b\sqrt{c+d\,x^2}\right]\partial_x\left(a\,x+b\sqrt{c+d\,x^2}\right)$$

Rule 1.3.3.10: If $a^2-b^2\,d==0\ \land\ b^2\,c+a==0$, then

$$\int \frac{\sqrt{a x^2 + b x \sqrt{c + d x^2}}}{x \sqrt{c + d x^2}} \, dx \to \frac{\sqrt{2} \, b}{a} \, \text{Subst}\left[\int \frac{1}{\sqrt{1 + \frac{x^2}{a}}} \, dx, \, x, \, a x + b \sqrt{c + d x^2}\right]$$

## Program code:

```
Int[Sqrt[a_.*x_^2+b_.*x_*Sqrt[c_+d_.*x_^2]]/(x_*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Sqrt[2]*b/a*Subst[Int[1/Sqrt[1+x^2/a],x],x,a*x+b*Sqrt[c+d*x^2]] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2*d,0] && EqQ[b^2*c+a,0]
```

**11:** $\int \frac{\sqrt{e x \left(a x + b \sqrt{c + d x^2}\right)}}{x \sqrt{c + d x^2}} \, dx$ when $a^2 - b^2 d == 0 \land b^2 c e + a == 0$

## Derivation: Algebraic normalization

## Rule 1.3.3.11: If $a^2 - b^2 d == 0 \land b^2 c e + a == 0$, then

$$\int \frac{\sqrt{e x \left(a x + b \sqrt{c + d x^2}\right)}}{x \sqrt{c + d x^2}} \, dx \to \int \frac{\sqrt{a e x^2 + b e x \sqrt{c + d x^2}}}{x \sqrt{c + d x^2}} \, dx$$

## Program code:

```
Int[Sqrt[e_.*x_*(a_.*x_+b_.*Sqrt[c_+d_.*x_^2])]/(x_*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Int[Sqrt[a*e*x^2+b*e*x*Sqrt[c+d*x^2]]/(x*Sqrt[c+d*x^2]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[a^2-b^2*d,0] && EqQ[b^2*c*e+a,0]
```

**12.** $\int \frac{u \sqrt{c x^2 + d \sqrt{a + b x^4}}}{\sqrt{a + b x^4}} \, dx$

**1:** $\int \frac{\sqrt{c\,x^2 + d\,\sqrt{a + b\,x^4}}}{\sqrt{a + b\,x^4}}\,dx$ when $c^2 - b\,d^2 == 0$

## Derivation: Integration by substitution

- Basis: If $c^2 - b\,d^2 == 0$, then $\frac{\sqrt{c\,x^2+d\,\sqrt{a+b\,x^4}}}{\sqrt{a+b\,x^4}} == d\,\text{Subst}\left[\frac{1}{1-2\,c\,x^2}, x, \frac{x}{\sqrt{c\,x^2+d\,\sqrt{a+b\,x^4}}}\right]\,\partial_x \frac{x}{\sqrt{c\,x^2+d\,\sqrt{a+b\,x^4}}}$

## Rule 1.3.3.12.1: If $c^2 - b\,d^2 == 0$, then

$$\int \frac{\sqrt{c\,x^2 + d\,\sqrt{a + b\,x^4}}}{\sqrt{a + b\,x^4}}\,dx \longrightarrow d\,\text{Subst}\left[\int \frac{1}{1 - 2\,c\,x^2}\,dx, x, \frac{x}{\sqrt{c\,x^2 + d\,\sqrt{a + b\,x^4}}}\right]$$

## Program code:

```
Int[Sqrt[c_.*x_^2+d_.*Sqrt[a_+b_.*x_^4]]/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  d*Subst[Int[1/(1-2*c*x^2),x],x,x/Sqrt[c*x^2+d*Sqrt[a+b*x^4]]] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2-b*d^2,0]
```

**2:** $\int \frac{(c+d\,x)^m \sqrt{b\,x^2+\sqrt{a+b^2\,x^4}}}{\sqrt{a+b^2\,x^4}}\,dx$

Author: Martin Welz on the sci.math.symbolic Usenet group

Derivation: Algebraic expansion

- Basis: If $a>0$, then $\sqrt{a+z^2} = \sqrt{\sqrt{a}-i\,z}\,\sqrt{\sqrt{a}+i\,z}$
- Basis: If $a>0$, then $\frac{\sqrt{z+\sqrt{a+z^2}}}{\sqrt{a+z^2}} = \frac{1-i}{2\sqrt{\sqrt{a}-i\,z}} + \frac{1+i}{2\sqrt{\sqrt{a}+i\,z}}$

Rule 1.3.3.12.2: If $a>0$, then

$$\int \frac{(c+d\,x)^m \sqrt{b\,x^2+\sqrt{a+b^2\,x^4}}}{\sqrt{a+b^2\,x^4}}\,dx \rightarrow \frac{1-i}{2}\int \frac{(c+d\,x)^m}{\sqrt{\sqrt{a}-i\,b\,x^2}}\,dx + \frac{1+i}{2}\int \frac{(c+d\,x)^m}{\sqrt{\sqrt{a}+i\,b\,x^2}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sqrt[b_.*x_^2+Sqrt[a_+e_.*x_^4]]/Sqrt[a_+e_.*x_^4],x_Symbol] :=
  (1-I)/2*Int[(c+d*x)^m/Sqrt[Sqrt[a]-I*b*x^2],x] +
  (1+I)/2*Int[(c+d*x)^m/Sqrt[Sqrt[a]+I*b*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[e,b^2] && GtQ[a,0]
```

**13.** $\int u\,(a+b\,x^3)^p\,dx$ when $p^2 = \frac{1}{4}$

**1.** $\int \frac{1}{(c+d\,x)\,\sqrt{a+b\,x^3}}\,dx$

1: $\int \frac{1}{(c+d\,x)\,\sqrt{a+b\,x^3}}\,dx$ when $b\,c^3-4\,a\,d^3==0$

## Derivation: Algebraic expansion

Basis: $\frac{1}{c+d\,x}==\frac{2}{3\,c}+\frac{c-2\,d\,x}{3\,c\,(c+d\,x)}$

Note: Second integrand is of the form $\frac{e+f\,x}{(c+d\,x)\,\sqrt{a+b\,x^3}}$ where $b\,c^3-4\,a\,d^3==0\wedge 2\,d\,e+c\,f==0$.

Rule 1.3.3.13.1.1: If $b\,c^3-4\,a\,d^3==0$, then

$$\int \frac{1}{(c+d\,x)\,\sqrt{a+b\,x^3}}\,dx \longrightarrow \frac{2}{3\,c}\int \frac{1}{\sqrt{a+b\,x^3}}\,dx+\frac{1}{3\,c}\int \frac{c-2\,d\,x}{(c+d\,x)\,\sqrt{a+b\,x^3}}\,dx$$

## Program code:

```
Int[1/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  2/(3*c)*Int[1/Sqrt[a+b*x^3],x] + 1/(3*c)*Int[(c-2*d*x)/((c+d*x)*Sqrt[a+b*x^3]),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c^3-4*a*d^3,0]
```

**2:** $\int \frac{1}{(c+d x) \sqrt{a+b x^3}} \, dx$ when $b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0$

## Derivation: Algebraic expansion

Basis: $\frac{1}{c+d x} == \frac{1}{c (3-z)} + \frac{c (2-z) - d x}{c (3-z) (c+d x)}$

Basis: $\frac{1}{c+d x} == -\frac{6 a d^3}{c \left(b c^3 - 28 a d^3\right)} + \frac{c \left(b c^3 - 22 a d^3\right) + 6 a d^4 x}{c \left(b c^3 - 28 a d^3\right) (c+d x)}$

Note: Second integrand is of the form $\frac{e+f x}{(c+d x) \sqrt{a+b x^3}}$ where $b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) == 0$.

Rule 1.3.3.13.1.2: If $b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0$, then

$$\int \frac{1}{(c+d x) \sqrt{a+b x^3}} \, dx \to -\frac{6 a d^3}{c \left(b c^3 - 28 a d^3\right)} \int \frac{1}{\sqrt{a+b x^3}} \, dx + \frac{1}{c \left(b c^3 - 28 a d^3\right)} \int \frac{c \left(b c^3 - 22 a d^3\right) + 6 a d^4 x}{(c+d x) \sqrt{a+b x^3}} \, dx$$

## Program code:

```
Int[1/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  -6*a*d^3/(c*(b*c^3-28*a*d^3))*Int[1/Sqrt[a+b*x^3],x] +
  1/(c*(b*c^3-28*a*d^3))*Int[Simp[c*(b*c^3-22*a*d^3)+6*a*d^4*x,x]/((c+d*x)*Sqrt[a+b*x^3]),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2*c^6-20*a*b*c^3*d^3-8*a^2*d^6,0]
```

**3:** $\int \frac{1}{(c+d\,x)\sqrt{a+b\,x^3}}\,dx$ when $b^2\,c^6-20\,a\,b\,c^3\,d^3-8\,a^2\,d^6\neq 0$

## Derivation: Algebraic expansion

Basis: $\frac{1}{c+d\,x} = -\frac{q}{\left(1+\sqrt{3}\right)d-c\,q}+\frac{d\left(1+\sqrt{3}+q\,x\right)}{\left(\left(1+\sqrt{3}\right)d-c\,q\right)(c+d\,x)}$

Note: Second integrand is of the form $\frac{e+f\,x}{(c+d\,x)\sqrt{a+b\,x^3}}$ where $b^2\,e^6-20\,a\,b\,e^3\,f^3-8\,a^2\,f^6 == 0$.

Rule 1.3.3.13.1.3: If $b^2\,c^6-20\,a\,b\,c^3\,d^3-8\,a^2\,d^6\neq 0$, let $q\to\left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{1}{(c+d\,x)\sqrt{a+b\,x^3}}\,dx \longrightarrow -\frac{q}{\left(1+\sqrt{3}\right)d-c\,q}\int\frac{1}{\sqrt{a+b\,x^3}}\,dx+\frac{d}{\left(1+\sqrt{3}\right)d-c\,q}\int\frac{1+\sqrt{3}+q\,x}{(c+d\,x)\sqrt{a+b\,x^3}}\,dx$$

## Program code:

```
Int[1/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{q=Rt[b/a,3]},
  -q/((1+Sqrt[3])*d-c*q)*Int[1/Sqrt[a+b*x^3],x] +
  d/((1+Sqrt[3])*d-c*q)*Int[(1+Sqrt[3]+q*x)/((c+d*x)*Sqrt[a+b*x^3]),x]] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2*c^6-20*a*b*c^3*d^3-8*a^2*d^6,0]
```

2. $\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} dx$ when $d e - c f \neq 0$

   1. $\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} dx$ when $d e - c f \neq 0 \land (b c^3 - 4 a d^3 == 0 \lor b c^3 + 8 a d^3 == 0)$

      1. $\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} dx$ when $d e - c f \neq 0 \land (b c^3 - 4 a d^3 == 0 \lor b c^3 + 8 a d^3 == 0) \land 2 d e + c f == 0$

         1: $\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} dx$ when $d e - c f \neq 0 \land b c^3 - 4 a d^3 == 0 \land 2 d e + c f == 0$

## Derivation: Integration by substitution

Basis: If $b c^3 - 4 a d^3 == 0 \land 2 d e + c f == 0$, then $\frac{e+f x}{(c+d x) \sqrt{a+b x^3}} == \frac{2 e}{d} \text{Subst}\left[\frac{1}{1+3 a x^2}, x, \frac{1+\frac{2 d x}{c}}{\sqrt{a+b x^3}}\right] \partial_x \frac{1+\frac{2 d x}{c}}{\sqrt{a+b x^3}}$

Rule 1.3.3.13.2.1.1.1: If $d e - c f \neq 0 \land b c^3 - 4 a d^3 == 0 \land 2 d e + c f == 0$, then

$$\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} dx \rightarrow \frac{2 e}{d} \text{Subst}\left[\int \frac{1}{1+3 a x^2} dx, x, \frac{1+\frac{2 d x}{c}}{\sqrt{a+b x^3}}\right]$$

## Program code:

```
Int[(e_+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  2*e/d*Subst[Int[1/(1+3*a*x^2),x],x,(1+2*d*x/c)/Sqrt[a+b*x^3]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && EqQ[b*c^3-4*a*d^3,0] && EqQ[2*d*e+c*f,0]
```

**2:** $\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} \, dx$ when $d e-c f \neq 0 \land b c^3+8 a d^3 == 0 \land 2 d e+c f == 0$

## Derivation: Integration by substitution

- Basis: If $b c^3+8 a d^3 == 0 \land 2 d e+c f == 0$, then $\frac{e+f x}{(c+d x) \sqrt{a+b x^3}} == -\frac{2 e}{d} \text{Subst}\left[\frac{1}{9-a x^2}, x, \frac{\left(1+\frac{f x}{e}\right)^2}{\sqrt{a+b x^3}}\right] \partial_x \frac{\left(1+\frac{f x}{e}\right)^2}{\sqrt{a+b x^3}}$

Rule 1.3.3.13.2.1.1.2: If $d e-c f \neq 0 \land b c^3+8 a d^3 == 0 \land 2 d e+c f == 0$, then

$$\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} \, dx \to -\frac{2 e}{d} \text{Subst}\left[\int \frac{1}{9-a x^2} \, dx, x, \frac{\left(1+\frac{f x}{e}\right)^2}{\sqrt{a+b x^3}}\right]$$

## Program code:

```
Int[(e_+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  -2*e/d*Subst[Int[1/(9-a*x^2),x],x,(1+f*x/e)^2/Sqrt[a+b*x^3]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && EqQ[b*c^3+8*a*d^3,0] && EqQ[2*d*e+c*f,0]
```

**2:** $\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} \, dx$ when $d e-c f \neq 0 \land \left(b c^3-4 a d^3 == 0 \lor b c^3+8 a d^3 == 0\right) \land 2 d e+c f \neq 0$

## Derivation: Algebraic expansion

- Basis: $\frac{e+f x}{c+d x} == \frac{2 d e+c f}{3 c d}+\frac{(d e-c f) (c-2 d x)}{3 c d (c+d x)}$

Note: Second integrand is of the form $\frac{e+f x}{(c+d x) \sqrt{a+b x^3}}$ where $\left(b c^3-4 a d^3 == 0 \lor b c^3+8 a d^3 == 0\right) \land 2 d e+c f == 0$.

Rule 1.3.3.13.2.1.2: If $d e-c f \neq 0 \land \left(b c^3-4 a d^3 == 0 \lor b c^3+8 a d^3 == 0\right) \land 2 d e+c f \neq 0$, then

$$\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} \, dx \longrightarrow \frac{2 d e+c f}{3 c d} \int \frac{1}{\sqrt{a+b x^3}} \, dx+\frac{d e-c f}{3 c d} \int \frac{c-2 d x}{(c+d x) \sqrt{a+b x^3}} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  (2*d*e+c*f)/(3*c*d)*Int[1/Sqrt[a+b*x^3],x] +
  (d*e-c*f)/(3*c*d)*Int[(c-2*d*x)/((c+d*x)*Sqrt[a+b*x^3]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && (EqQ[b*c^3-4*a*d^3,0] || EqQ[b*c^3+8*a*d^3,0]) && NeQ[2*d*e+c*f,0]
```

2. $\int \frac{e + f x}{(c + d x) \sqrt{a + b x^3}} \, dx$ when $d e - c f \neq 0 \land b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0$

1: $\int \frac{e + f x}{(c + d x) \sqrt{a + b x^3}} \, dx$ when $d e - c f \neq 0 \land b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) == 0$

Derivation: Integration by substitution

Basis: If $b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) == 0$, let $k \to \frac{d e + 2 c f}{c f}$, then

$$\frac{e + f x}{(c + d x) \sqrt{a + b x^3}} == \frac{(1 + k) e}{d} \text{Subst}\left[\frac{1}{1 + (3 + 2 k) a x^2}, x, \frac{1 + \frac{(1+k) d x}{c}}{\sqrt{a + b x^3}}\right] \partial_x \frac{1 + \frac{(1+k) d x}{c}}{\sqrt{a + b x^3}}$$

Note: If $b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) == 0$, then $d^2 e^2 + 4 c d e f + c^2 f^2 == 0$, so $\frac{d e + 2 c f}{c f}$ must equal $\sqrt{3}$ or $-\sqrt{3}$.

Rule 1.3.3.13.2.2.1: If $d e - c f \neq 0 \land b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) == 0$, let $k \to \frac{d e + 2 c f}{c f}$, then

$$\int \frac{e + f x}{(c + d x) \sqrt{a + b x^3}} \, dx \to \frac{(1 + k) e}{d} \text{Subst}\left[\int \frac{1}{1 + (3 + 2 k) a x^2} \, dx, x, \frac{1 + \frac{(1+k) d x}{c}}{\sqrt{a + b x^3}}\right]$$

Program code:

```
Int[(e_+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{k=Simplify[(d*e+2*c*f)/(c*f)]},
  (1+k)*e/d*Subst[Int[1/(1+(3+2*k)*a*x^2),x],x,(1+(1+k)*d*x/c)/Sqrt[a+b*x^3]]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && EqQ[b^2*c^6-20*a*b*c^3*d^3-8*a^2*d^6,0] && EqQ[6*a*d^4*e-c*f*(b*c^3-22*a*d^3),0]
```

**2:** $\int \frac{e + f x}{(c + d x) \sqrt{a + b x^3}} \, dx$ when $d e - c f \neq 0 \land b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{e+f x}{c+d x} == \frac{d e+(2-z) c f}{c d (3-z)} + \frac{(d e-c f) ((2-z) c-d x)}{c d (3-z) (c+d x)}$

Basis: $\frac{e+f x}{c+d x} == -\frac{6 a d^4 e-c \left(b c^3-22 a d^3\right) f}{c d \left(b c^3-28 a d^3\right)} + \frac{(d e-c f) \left(c \left(b c^3-22 a d^3\right)+6 a d^4 x\right)}{c d \left(b c^3-28 a d^3\right) (c+d x)}$

Note: Second integrand is of the form $\frac{e+f x}{(c+d x) \sqrt{a+b x^3}}$ where $b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) == 0$.

Rule 1.3.3.13.2.2.2: If $d e - c f \neq 0 \land b^2 c^6 - 20 a b c^3 d^3 - 8 a^2 d^6 == 0 \land 6 a d^4 e - c f \left(b c^3 - 22 a d^3\right) \neq 0$, then

$$\int \frac{e + f x}{(c + d x) \sqrt{a + b x^3}} \, dx \to -\frac{6 a d^4 e - c f \left(b c^3 - 22 a d^3\right)}{c d \left(b c^3 - 28 a d^3\right)} \int \frac{1}{\sqrt{a + b x^3}} \, dx + \frac{d e - c f}{c d \left(b c^3 - 28 a d^3\right)} \int \frac{c \left(b c^3 - 22 a d^3\right) + 6 a d^4 x}{(c + d x) \sqrt{a + b x^3}} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  -(6*a*d^4*e-c*f*(b*c^3-22*a*d^3))/(c*d*(b*c^3-28*a*d^3))*Int[1/Sqrt[a+b*x^3],x] +
  (d*e-c*f)/(c*d*(b*c^3-28*a*d^3))*Int[(c*(b*c^3-22*a*d^3)+6*a*d^4*x)/((c+d*x)*Sqrt[a+b*x^3]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && EqQ[b^2*c^6-20*a*b*c^3*d^3-8*a^2*d^6,0] && NeQ[6*a*d^4*e-c*f*(b*c^3-22*a*d^3),0]
```

3. $\int \frac{e+f x}{(c+d x)\sqrt{a+b x^3}}\,dx$ when $d e-c f\neq 0\land b^2 e^6-20 a b e^3 f^3-8 a^2 f^6==0$

1: $\int \frac{e+f x}{(c+d x)\sqrt{a+b x^3}}\,dx$ when $d e-c f\neq 0\land b e^3-2\left(5+3\sqrt{3}\right) a f^3==0\land b c^3-2\left(5-3\sqrt{3}\right) a d^3\neq 0$

Reference: G&R 3.139

Derivation: Piecewise constant extraction and integration by substitution (the Möbius transformation)

- Basis: Let $q\to\left(\frac{b}{a}\right)^{1/3}$, then $\partial_x \frac{\left(1+\sqrt{3}+q x\right)^2\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}}{\sqrt{a+b x^3}}==0$

Basis:
$$\frac{1}{(c+d x)\left(1+\sqrt{3}+q x\right)\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}}==$$
$$4\times 3^{1/4}\sqrt{2-\sqrt{3}}\ \text{Subst}\left[1\Big/\left(\left(\left(1-\sqrt{3}\right) d-c q+\left(\left(1+\sqrt{3}\right) d-c q\right) x\right)\sqrt{\left(1-x^2\right)\left(7-4\sqrt{3}+x^2\right)}\right),\ x,\ \frac{-1+\sqrt{3}-q x}{1+\sqrt{3}+q x}\right]\partial_x \frac{-1+\sqrt{3}-q x}{1+\sqrt{3}+q x}$$

- Basis: $\sqrt{\left(1-x^2\right)\left(7-4\sqrt{3}+x^2\right)}==\sqrt{1-x^2}\sqrt{7-4\sqrt{3}+x^2}$

Rule 1.3.3.13.2.3.1: If $d e-c f\neq 0\land b e^3-2\left(5+3\sqrt{3}\right) a f^3==0\land b c^3-2\left(5-3\sqrt{3}\right) a d^3\neq 0$, let $q\to\left(\frac{b}{a}\right)^{1/3}\to\frac{\left(1+\sqrt{3}\right) f}{e}$, then

$$\int \frac{e+f x}{(c+d x)\sqrt{a+b x^3}}\,dx\to\frac{f\left(1+\sqrt{3}+q x\right)^2\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}}{q\sqrt{a+b x^3}}\int \frac{1}{(c+d x)\left(1+\sqrt{3}+q x\right)\sqrt{\frac{1+q^3 x^3}{\left(1+\sqrt{3}+q x\right)^4}}}\,dx$$

$$\to \frac{4\times 3^{1/4}\sqrt{2-\sqrt{3}}\ f\left(1+\sqrt{3}+q\,x\right)^2\sqrt{\frac{1+q^3\,x^3}{\left(1+\sqrt{3}+q\,x\right)^4}}}{q\sqrt{a+b\,x^3}}\ \text{Subst}\left[\int\frac{1}{\left(\left(1-\sqrt{3}\right)d-c\,q+\left(\left(1+\sqrt{3}\right)d-c\,q\right)x\right)\sqrt{\left(1-x^2\right)\left(7-4\sqrt{3}+x^2\right)}}\,dx,\ x,\ \frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right]$$

$$\to \frac{4\times 3^{1/4}\sqrt{2-\sqrt{3}}\ f\,(1+q\,x)\sqrt{\frac{1-q\,x+q^2\,x^2}{\left(1+\sqrt{3}+q\,x\right)^2}}}{q\sqrt{a+b\,x^3}\sqrt{\frac{1+q\,x}{\left(1+\sqrt{3}+q\,x\right)^2}}}\ \text{Subst}\left[\int\frac{1}{\left(\left(1-\sqrt{3}\right)d-c\,q+\left(\left(1+\sqrt{3}\right)d-c\,q\right)x\right)\sqrt{1-x^2}\sqrt{7-4\sqrt{3}+x^2}}\,dx,\ x,\ \frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right]$$

## Program code:

```
(* Int[(e_+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{q=(1+Sqrt[3])*f/e},
  4*3^(1/4)*Sqrt[2-Sqrt[3]]*f*(1+Sqrt[3]+q*x)^2*Sqrt[(1+q^3*x^3)/(1+Sqrt[3]+q*x)^4]/(q*Sqrt[a+b*x^3])*
  Subst[Int[1/(((1-Sqrt[3])*d-c*q+((1+Sqrt[3])*d-c*q)*x)*
      Sqrt[7-4*Sqrt[3]-2*(3-2*Sqrt[3])*x^2-x^4]),x],x,(-1+Sqrt[3]-q*x)/(1+Sqrt[3]+q*x)]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && EqQ[b*e^3-2*(5+3*Sqrt[3])*a*f^3,0] && NeQ[b*c^3-2*(5-3*Sqrt[3])*a*d^3,0] *)
```

```
Int[(e_+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{q=Simplify[(1+Sqrt[3])*f/e]},
  4*3^(1/4)*Sqrt[2-Sqrt[3]]*f*(1+q*x)*Sqrt[(1-q*x+q^2*x^2)/(1+Sqrt[3]+q*x)^2]/
    (q*Sqrt[a+b*x^3]*Sqrt[(1+q*x)/(1+Sqrt[3]+q*x)^2])*
    Subst[Int[1/(((1-Sqrt[3])*d-c*q+((1+Sqrt[3])*d-c*q)*x)*Sqrt[1-x^2]*Sqrt[7-4*Sqrt[3]+x^2]),x],x,(-1+Sqrt[3]-q*x)/(1+Sqrt[3]+q*x)]]
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && EqQ[b*e^3-2*(5+3*Sqrt[3])*a*f^3,0] && NeQ[b*c^3-2*(5-3*Sqrt[3])*a*d^3,0]
```

**2:** $\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} \, dx$ when $d e-c f \neq 0 \land b e^3-2 \left(5-3 \sqrt{3}\right) a f^3 == 0 \land b c^3-2 \left(5+3 \sqrt{3}\right) a d^3 \neq 0$

Reference: G&R 3.139

Derivation: Piecewise constant extraction and integration by substitution (the Möbius transformation)

- Basis: Let $q \to \left(-\frac{b}{a}\right)^{1/3}$, then $\partial_x \frac{\left(1-\sqrt{3}-q x\right)^2 \sqrt{-\frac{1-q^3 x^3}{\left(1-\sqrt{3}-q x\right)^4}}}{\sqrt{a+b x^3}} == 0$

Basis:

$$\frac{1}{(c+d x) \left(1-\sqrt{3}-q x\right) \sqrt{-\frac{1-q^3 x^3}{\left(1-\sqrt{3}-q x\right)^4}}} ==$$

$$4 \times 3^{1/4} \sqrt{2+\sqrt{3}} \; \text{Subst}\left[1 \Big/ \left(\left(\left(1+\sqrt{3}\right) d+c q+\left(\left(1-\sqrt{3}\right) d+c q\right) x\right) \sqrt{\left(1-x^2\right) \left(7+4 \sqrt{3}+x^2\right)}\right),\; x,\; \frac{1+\sqrt{3}-q x}{-1+\sqrt{3}+q x}\right] \partial_x \frac{1+\sqrt{3}-q x}{-1+\sqrt{3}+q x}$$

- Basis: $\sqrt{\left(1-x^2\right) \left(7+4 \sqrt{3}+x^2\right)} == \sqrt{1-x^2} \sqrt{7+4 \sqrt{3}+x^2}$

Rule 1.3.3.13.2.3.2: If $d e-c f \neq 0 \land b e^3-2 \left(5-3 \sqrt{3}\right) a f^3 == 0 \land b c^3-2 \left(5+3 \sqrt{3}\right) a d^3 \neq 0$, let $q \to \frac{\left(-1+\sqrt{3}\right) f}{e}$, then

$$\int \frac{e+f x}{(c+d x) \sqrt{a+b x^3}} \, dx \longrightarrow -\frac{f \left(1-\sqrt{3}-q x\right)^2 \sqrt{-\frac{1-q^3 x^3}{\left(1-\sqrt{3}-q x\right)^4}}}{q \sqrt{a+b x^3}} \int \frac{1}{(c+d x) \left(1-\sqrt{3}-q x\right) \sqrt{-\frac{1-q^3 x^3}{\left(1-\sqrt{3}-q x\right)^4}}} \, dx$$

$$\to -\frac{4\times 3^{1/4}\sqrt{2+\sqrt{3}}\ f\left(1-\sqrt{3}-q\,x\right)^2\sqrt{-\frac{1-q^3x^3}{\left(1-\sqrt{3}-q\,x\right)^4}}}{q\sqrt{a+b\,x^3}}\ \text{Subst}\left[\int\frac{1}{\left(\left(1+\sqrt{3}\right)d+c\,q+\left(\left(1-\sqrt{3}\right)d+c\,q\right)x\right)\sqrt{\left(1-x^2\right)\left(7+4\sqrt{3}+x^2\right)}}\,dx,\ x,\ \frac{1+\sqrt{3}-q\,x}{-1+\sqrt{3}+q\,x}\right]$$

$$\to \frac{4\times 3^{1/4}\sqrt{2+\sqrt{3}}\ f\,(1-q\,x)\sqrt{\frac{1+q\,x+q^2x^2}{\left(1-\sqrt{3}-q\,x\right)^2}}}{q\sqrt{a+b\,x^3}\sqrt{-\frac{1-q\,x}{\left(1-\sqrt{3}-q\,x\right)^2}}}\ \text{Subst}\left[\int\frac{1}{\left(\left(1+\sqrt{3}\right)d+c\,q+\left(\left(1-\sqrt{3}\right)d+c\,q\right)x\right)\sqrt{1-x^2}\sqrt{7+4\sqrt{3}+x^2}}\,dx,\ x,\ \frac{1+\sqrt{3}-q\,x}{-1+\sqrt{3}+q\,x}\right]$$

## Program code:

```
Int[(e_+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{q=Simplify[(-1+Sqrt[3])*f/e]},
  4*3^(1/4)*Sqrt[2+Sqrt[3]]*f*(1-q*x)*Sqrt[(1+q*x+q^2*x^2)/(1-Sqrt[3]-q*x)^2]/
    (q*Sqrt[a+b*x^3]*Sqrt[-(1-q*x)/(1-Sqrt[3]-q*x)^2])*
  Subst[Int[1/(((1+Sqrt[3])*d+c*q+((1-Sqrt[3])*d+c*q)*x)*Sqrt[1-x^2]*Sqrt[7+4*Sqrt[3]+x^2]),x],x,(1+Sqrt[3]-q*x)/(-1+Sqrt[3]+q*x)]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && EqQ[b*e^3-2*(5-3*Sqrt[3])*a*f^3,0] && NeQ[b*c^3-2*(5+3*Sqrt[3])*a*d^3,0]
```

**4:** $\int\frac{e+f\,x}{(c+d\,x)\sqrt{a+b\,x^3}}\,dx$ when $d\,e-c\,f\neq 0\ \wedge\ b^2c^6-20\,a\,b\,c^3d^3-8\,a^2d^6\neq 0\ \wedge\ b^2e^6-20\,a\,b\,e^3f^3-8\,a^2f^6\neq 0$

## Derivation: Algebraic expansion

Basis: $\frac{e+f\,x}{c+d\,x}=\frac{\left(1+\sqrt{3}\right)f-e\,q}{\left(1+\sqrt{3}\right)d-c\,q}+\frac{(d\,e-c\,f)\left(1+\sqrt{3}+q\,x\right)}{\left(\left(1+\sqrt{3}\right)d-c\,q\right)(c+d\,x)}$

Note: Second integrand is of the form $\frac{e+f\,x}{(c+d\,x)\sqrt{a+b\,x^3}}$ where $b^2e^6-20\,a\,b\,e^3f^3-8\,a^2f^6=0$.

Rule 1.3.3.13.2.4: If $d\,e-c\,f\neq 0\ \wedge\ b^2c^6-20\,a\,b\,c^3d^3-8\,a^2d^6\neq 0\ \wedge\ b^2e^6-20\,a\,b\,e^3f^3-8\,a^2f^6\neq 0$, let $q\to\left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{e + f x}{(c + d x) \sqrt{a + b x^3}} \, dx \longrightarrow \frac{\left(1 + \sqrt{3}\right) f - e q}{\left(1 + \sqrt{3}\right) d - c q} \int \frac{1}{\sqrt{a + b x^3}} \, dx + \frac{d e - c f}{\left(1 + \sqrt{3}\right) d - c q} \int \frac{1 + \sqrt{3} + q x}{(c + d x) \sqrt{a + b x^3}} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)/((c_+d_.*x_)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  With[{q=Rt[b/a,3]},
  ((1+Sqrt[3])*f-e*q)/((1+Sqrt[3])*d-c*q)*Int[1/Sqrt[a+b*x^3],x] +
  (d*e-c*f)/((1+Sqrt[3])*d-c*q)*Int[(1+Sqrt[3]+q*x)/((c+d*x)*Sqrt[a+b*x^3]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && NeQ[b^2*c^6-20*a*b*c^3*d^3-8*a^2*d^6,0] && NeQ[b^2*e^6-20*a*b*e^3*f^3-8*a^2*f^6,0]
```

3: $\int \frac{f + g x + h x^2}{\left(c + d x + e x^2\right) \sqrt{a + b x^3}} \, dx$ when $b d f - 2 a e h \neq 0 \land b g^3 - 8 a h^3 == 0 \land g^2 + 2 f h == 0 \land b d f + b c g - 4 a e h == 0$

## Derivation: Integration by substitution

## Basis: If $b g^3 - 8 a h^3 == 0 \land g^2 + 2 f h == 0 \land b d f + b c g - 4 a e h == 0$, then

$$\frac{f + g x + h x^2}{\left(c + d x + e x^2\right) \sqrt{a + b x^3}} == -2 g h \, \text{Subst}\left[\frac{1}{2 e h - (b d f - 2 a e h) x^2}, x, \frac{1 + \frac{2 h x}{g}}{\sqrt{a + b x^3}}\right] \partial_x \frac{1 + \frac{2 h x}{g}}{\sqrt{a + b x^3}}$$

## Rule 1.3.3.13.3: If $b d f - 2 a e h \neq 0 \land b g^3 - 8 a h^3 == 0 \land g^2 + 2 f h == 0 \land b d f + b c g - 4 a e h == 0$, then

$$\int \frac{f + g x + h x^2}{\left(c + d x + e x^2\right) \sqrt{a + b x^3}} \, dx \longrightarrow -2 g h \, \text{Subst}\left[\int \frac{1}{2 e h - (b d f - 2 a e h) x^2} \, dx, x, \frac{1 + \frac{2 h x}{g}}{\sqrt{a + b x^3}}\right]$$

## Program code:

```
Int[(f_+g_.*x_+h_.*x_^2)/((c_+d_.*x_+e_.*x_^2)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  -2*g*h*Subst[Int[1/(2*e*h-(b*d*f-2*a*e*h)*x^2),x],x,(1+2*h*x/g)/Sqrt[a+b*x^3]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && NeQ[b*d*f-2*a*e*h,0] && EqQ[b*g^3-8*a*h^3,0] && EqQ[g^2+2*f*h,0] && EqQ[b*d*f+b*c*g-4*a*e*h,0]
```

```
Int[(f_+g_.*x_+h_.*x_^2)/((c_+e_.*x_^2)*Sqrt[a_+b_.*x_^3]),x_Symbol] :=
  -g/e*Subst[Int[1/(1+a*x^2),x],x,(1+2*h*x/g)/Sqrt[a+b*x^3]] /;
FreeQ[{a,b,c,e,f,g,h},x] && EqQ[b*g^3-8*a*h^3,0] && EqQ[g^2+2*f*h,0] && EqQ[b*c*g-4*a*e*h,0]
```

4: $\int \frac{\sqrt{a+b x^3}}{c+d x} \, dx$

## Derivation: Algebraic expansion

## Basis: $\frac{\sqrt{a+b x^3}}{c+d x} == \frac{b x^2}{d \sqrt{a+b x^3}} - \frac{b c^3-a d^3}{d^3 (c+d x) \sqrt{a+b x^3}} + \frac{b c (c-d x)}{d^3 \sqrt{a+b x^3}}$

## Rule 1.3.3.13.4:

$$\int \frac{\sqrt{a+b x^3}}{c+d x} \, dx \to \frac{b}{d} \int \frac{x^2}{\sqrt{a+b x^3}} \, dx - \frac{b c^3 - a d^3}{d^3} \int \frac{1}{(c+d x) \sqrt{a+b x^3}} \, dx + \frac{b c}{d^3} \int \frac{c-d x}{\sqrt{a+b x^3}} \, dx$$

## Program code:

```
Int[Sqrt[a_+b_.*x_^3]/(c_+d_.*x_),x_Symbol] :=
  b/d*Int[x^2/Sqrt[a+b*x^3],x] -
  (b*c^3-a*d^3)/d^3*Int[1/((c+d*x)*Sqrt[a+b*x^3]),x] +
  b*c/d^3*Int[(c-d*x)/Sqrt[a+b*x^3],x] /;
FreeQ[{a,b,c,d},x]
```

**14.** $\int \frac{u}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx$

**1.** $\int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx$

**1:** $\int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx$ when $b\,c^3+a\,d^3 == 0$

## Rule 1.3.3.14.1.1: If $b\,c^3+a\,d^3 == 0$, then

$$\int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx \longrightarrow \frac{\sqrt{3}\,\mathrm{ArcTan}\left[\frac{1-\frac{2^{1/3}\,b^{1/3}\,(c-d\,x)}{d\,(a+b\,x^3)^{1/3}}}{\sqrt{3}}\right]}{2^{4/3}\,b^{1/3}\,c}+\frac{\mathrm{Log}\left[(c+d\,x)^2\,(c-d\,x)\right]}{2^{7/3}\,b^{1/3}\,c}-\frac{3\,\mathrm{Log}\left[b^{1/3}\,(c-d\,x)+2^{2/3}\,d\,(a+b\,x^3)^{1/3}\right]}{2^{7/3}\,b^{1/3}\,c}$$

## Program code:

```
Int[1/((c_+d_.*x_)*(a_+b_.*x_^3)^(1/3)),x_Symbol] :=
  Sqrt[3]*ArcTan[(1-2^(1/3)*Rt[b,3]*(c-d*x)/(d*(a+b*x^3)^(1/3)))/Sqrt[3]]/(2^(4/3)*Rt[b,3]*c) +
  Log[(c+d*x)^2*(c-d*x)]/(2^(7/3)*Rt[b,3]*c) -
  (3*Log[Rt[b,3]*(c-d*x)+2^(2/3)*d*(a+b*x^3)^(1/3)])/(2^(7/3)*Rt[b,3]*c) /;
FreeQ[{a,b,c,d},x] && EqQ[b*c^3+a*d^3,0]
```

**2:** $\int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx$ when $2\,b\,c^3-a\,d^3==0$

## Derivation: Algebraic expansion

## Basis: $\frac{1}{c+d\,x}==\frac{1}{2\,c}+\frac{c-d\,x}{2\,c\,(c+d\,x)}$

## Rule 1.3.3.14.1.2: If $2\,b\,c^3-a\,d^3==0$, then

$$\int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx \to \frac{1}{2\,c}\int \frac{1}{(a+b\,x^3)^{1/3}}\,dx+\frac{1}{2\,c}\int \frac{c-d\,x}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx$$

## Program code:

```
Int[1/((c_+d_.*x_)*(a_+b_.*x_^3)^(1/3)),x_Symbol] :=
  1/(2*c)*Int[1/(a+b*x^3)^(1/3),x] + 1/(2*c)*Int[(c-d*x)/((c+d*x)*(a+b*x^3)^(1/3)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[2*b*c^3-a*d^3,0]
```

**U:** $\int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx$

## Rule 1.3.3.14.1.U:

$$\int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx \to \int \frac{1}{(c+d\,x)\,(a+b\,x^3)^{1/3}}\,dx$$

## Program code:

```
Int[1/((c_+d_.*x_)*(a_+b_.*x_^3)^(1/3)),x_Symbol] :=
  Unintegrable[1/((c+d*x)*(a+b*x^3)^(1/3)),x] /;
FreeQ[{a,b,c,d},x]
```

**2.** $\int \frac{e+f x}{(c+d x)\left(a+b x^3\right)^{1/3}} \, dx$

**1:** $\int \frac{e+f x}{(c+d x)\left(a+b x^3\right)^{1/3}} \, dx$ when $d e+c f==0 \wedge 2 b c^3-a d^3==0$

## Rule 1.3.3.14.2.1: If $d e+c f==0 \wedge 2 b c^3-a d^3==0$, then

$$\int \frac{e+f x}{(c+d x)\left(a+b x^3\right)^{1/3}} \, dx \longrightarrow \frac{\sqrt{3}\, f \operatorname{ArcTan}\left[\frac{1+\frac{2 b^{1/3}(2 c+d x)}{d\left(a+b x^3\right)^{1/3}}}{\sqrt{3}}\right]}{b^{1/3} d}+\frac{f \operatorname{Log}[c+d x]}{b^{1/3} d}-\frac{3 f \operatorname{Log}\left[b^{1/3}(2 c+d x)-d\left(a+b x^3\right)^{1/3}\right]}{2 b^{1/3} d}$$

## Program code:

```
Int[(e_+f_.*x_)/((c_+d_.*x_)*(a_+b_.*x_^3)^(1/3)),x_Symbol] :=
  Sqrt[3]*f*ArcTan[(1+2*Rt[b,3]*(2*c+d*x)/(d*(a+b*x^3)^(1/3)))/Sqrt[3]]/(Rt[b,3]*d) +
  (f*Log[c+d*x])/(Rt[b,3]*d) -
  (3*f*Log[Rt[b,3]*(2*c+d*x)-d*(a+b*x^3)^(1/3)])/(2*Rt[b,3]*d) /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[d*e+c*f,0] && EqQ[2*b*c^3-a*d^3,0]
```

**2:** $\int \frac{e+f\,x}{(c+d\,x)\,\left(a+b\,x^3\right)^{1/3}}\,dx$

## Derivation: Algebraic expansion

## Basis: $\frac{e+f\,x}{c+d\,x} == \frac{f}{d} + \frac{d\,e-c\,f}{d\,(c+d\,x)}$

## Rule 1.3.3.14.2.2:

$$\int \frac{e+f\,x}{(c+d\,x)\,\left(a+b\,x^3\right)^{1/3}}\,dx \longrightarrow \frac{f}{d}\int \frac{1}{\left(a+b\,x^3\right)^{1/3}}\,dx + \frac{d\,e-c\,f}{d}\int \frac{1}{(c+d\,x)\,\left(a+b\,x^3\right)^{1/3}}\,dx$$

## Program code:

```
Int[(e_.+f_.*x_)/((c_.+d_.*x_)*(a_+b_.*x_^3)^(1/3)),x_Symbol] :=
  f/d*Int[1/(a+b*x^3)^(1/3),x] + (d*e-c*f)/d*Int[1/((c+d*x)*(a+b*x^3)^(1/3)),x] /;
FreeQ[{a,b,c,d,e,f},x]
```

15. $\int u\,(c+d\,x^n)^q\,(a+b\,x^{nn})^p\,dx$ when $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-\land\,\text{Log}\left[2,\frac{nn}{n}\right]\in\mathbb{Z}^+$

1: $\int (c+d\,x^n)^q\,(a+b\,x^{nn})^p\,dx$ when $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-\land\,\text{Log}\left[2,\frac{nn}{n}\right]\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: If $q\in\mathbb{Z}$, then $(c+d\,x^n)^q = \left(\frac{c}{c^2-d^2\,x^{2\,n}}-\frac{d\,x^n}{c^2-d^2\,x^{2\,n}}\right)^{-q}$

Note: Resulting integrands are of the form $x^m\,(a+b\,x^{nn})^p\,(c+d\,x^{2\,n})^q$ which are integrable in terms of the Appell hypergeometric function .

Rule 1.3.3.15.1: If $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-\land\,\text{Log}\left[2,\frac{nn}{n}\right]\in\mathbb{Z}^+$, then

$$\int (c+d\,x^n)^q\,(a+b\,x^{nn})^p\,dx\;\longrightarrow\;\int (a+b\,x^{nn})^p\,\text{ExpandIntegrand}\left[\left(\frac{c}{c^2-d^2\,x^{2\,n}}-\frac{d\,x^n}{c^2-d^2\,x^{2\,n}}\right)^{-q},\,x\right]dx$$

Program code:

```
Int[(c_+d_.*x_^n_.)^q_*(a_+b_.*x_^nn_.)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^nn)^p,(c/(c^2-d^2*x^(2*n))-d*x^n/(c^2-d^2*x^(2*n)))^(-q),x],x] /;
FreeQ[{a,b,c,d,n,nn,p},x] && Not[IntegerQ[p]] && ILtQ[q,0] && IGtQ[Log[2,nn/n],0]
```

2: $\int (e\,x)^m\,(c+d\,x^n)^q\,(a+b\,x^{nn})^p\,dx$ when $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-\land\,\text{Log}\left[2,\frac{nn}{n}\right]\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: If $q\in\mathbb{Z}$, then $(c+d\,x^n)^q = \left(\frac{c}{c^2-d^2\,x^{2\,n}}-\frac{d\,x^n}{c^2-d^2\,x^{2\,n}}\right)^{-q}$

Note: Resulting integrands are of the form $x^m\,(a+b\,x^{nn})^p\,(c+d\,x^{2\,n})^q$ which are integrable in terms of the Appell hypergeometric function .

Rule 1.3.3.15.2.1: If $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-\land\,\text{Log}\left[2,\frac{nn}{n}\right]\in\mathbb{Z}^+$, then

$$\int (e\,x)^m\,(c+d\,x^n)^q\,(a+b\,x^{nn})^p\,dx \to \frac{(e\,x)^m}{x^m}\int x^m\,(a+b\,x^{nn})^p\,\text{ExpandIntegrand}\left[\left(\frac{c}{c^2-d^2\,x^{2\,n}}-\frac{d\,x^n}{c^2-d^2\,x^{2\,n}}\right)^{-q},x\right]dx$$

## Program code:

```
Int[(e_.*x_)^m_.*(c_+d_.*x_^n_.)^q_*(a_+b_.*x_^nn_.)^p_,x_Symbol] :=
  (e*x)^m/x^m*Int[ExpandIntegrand[x^m*(a+b*x^nn)^p,(c/(c^2-d^2*x^(2*n))-d*x^n/(c^2-d^2*x^(2*n)))^(-q),x],x] /;
FreeQ[{a,b,c,d,e,m,n,nn,p},x] && Not[IntegerQ[p]] && ILtQ[q,0] && IGtQ[Log[2,nn/n],0]
```

16. $\int \frac{u}{c+d\,x^n+e\,\sqrt{a+b\,x^n}}\,dx$ when $b\,c-a\,d==0$

1: $\int \frac{x^m}{c+d\,x^n+e\,\sqrt{a+b\,x^n}}\,dx$ when $b\,c-a\,d==0\,\land\,\frac{m+1}{n}\in\mathbb{Z}$

## Derivation: Integration by substitution

## Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F[x^n]==\frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],x,x^n\right]\partial_x x^n$

## Rule 1.3.3.16.1: If $b\,c-a\,d==0\,\land\,\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int \frac{x^m}{c+d\,x^n+e\,\sqrt{a+b\,x^n}}\,dx \to \frac{1}{n}\,\text{Subst}\left[\int \frac{x^{\frac{m+1}{n}-1}}{c+d\,x+e\,\sqrt{a+b\,x}}\,dx,x,x^n\right]$$

## Program code:

```
Int[x_^m_./(c_+d_.*x_^n_+e_.*Sqrt[a_+b_.*x_^n_]),x_Symbol] :=
  1/n*Subst[Int[x^((m+1)/n-1)/(c+d*x+e*Sqrt[a+b*x]),x],x,x^n] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[b*c-a*d,0] && IntegerQ[(m+1)/n]
```

2: $\int \frac{u}{c + d x^n + e \sqrt{a + b x^n}} \, dx$ when $b c - a d == 0$

## Derivation: Algebraic expansion

Basis: If $b c - a d == 0$, then $\frac{1}{c + d z + e \sqrt{a + b z}} == \frac{c}{c^2 - a e^2 + c d z} - \frac{a e}{\left(c^2 - a e^2 + c d z\right) \sqrt{a + b z}}$

Rule 1.3.3.16.2: If $b c - a d == 0$, then

$$\int \frac{u}{c + d x^n + e \sqrt{a + b x^n}} \, dx \longrightarrow c \int \frac{u}{c^2 - a e^2 + c d x^n} \, dx - a e \int \frac{u}{\left(c^2 - a e^2 + c d x^n\right) \sqrt{a + b x^n}} \, dx$$

## Program code:

```
Int[u_./(c_+d_.*x_^n_+e_.*Sqrt[a_+b_.*x_^n_]),x_Symbol] :=
  c*Int[u/(c^2-a*e^2+c*d*x^n),x] - a*e*Int[u/((c^2-a*e^2+c*d*x^n)*Sqrt[a+b*x^n]),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[b*c-a*d,0]
```