# Rules for integrands of the form $(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n$

1. $\int \text{Cos}[e+f\,x]^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $\frac{p-1}{2}\in\mathbb{Z}$

   1: $\int \text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$

## Derivation: Integration by substitution

Basis: $\text{Cos}[e+f\,x]\,F[\text{Sin}[e+f\,x]] = \frac{1}{b\,f}\,\text{Subst}\left[F\left[\frac{x}{b}\right],\,x,\,b\,\text{Sin}[e+f\,x]\right]\,\partial_x\,(b\,\text{Sin}[e+f\,x])$

Rule:

$$\int \text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx \longrightarrow \frac{1}{b\,f}\,\text{Subst}\left[\int (a+x)^m\left(c+\frac{d}{b}\,x\right)^n dx,\,x,\,b\,\text{Sin}[e+f\,x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]*(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  1/(b*f)*Subst[Int[(a+x)^m*(c+d/b*x)^n,x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

**2:** $\int \cos[e+f x]^p (d \sin[e+f x])^n (a+b \sin[e+f x]) \, dx$ when $\frac{p-1}{2} \in \mathbb{Z} \land n \in \mathbb{Z} \land (p < 0 \land a^2-b^2 \neq 0 \lor 0 < n < p-1 \lor p+1 < -n < 2 p+1)$

Derivation: Algebraic expansion

Rule: If $\frac{p-1}{2} \in \mathbb{Z} \land n \in \mathbb{Z} \land (p < 0 \land a^2-b^2 \neq 0 \lor 0 < n < p-1 \lor p+1 < -n < 2 p+1)$, then

$$\int \cos[e+f x]^p (d \sin[e+f x])^n (a+b \sin[e+f x]) \, dx \rightarrow a \int \cos[e+f x]^p (d \sin[e+f x])^n \, dx + \frac{b}{d} \int \cos[e+f x]^p (d \sin[e+f x])^{n+1} \, dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_*(d_.*sin[e_.+f_.*x_])^n_.*(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  a*Int[Cos[e+f*x]^p*(d*Sin[e+f*x])^n,x] + b/d*Int[Cos[e+f*x]^p*(d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,n,p},x] && IntegerQ[(p-1)/2] && IntegerQ[n] && (LtQ[p,0] && NeQ[a^2-b^2,0] || LtQ[0,n,p-1] || LtQ[p+1,-n,2*p+1])
```

3. $\int \cos[e+f x]^p \, (a+b \sin[e+f x])^m \, (c+d \sin[e+f x])^n \, dx$ when $\frac{p-1}{2} \in \mathbb{Z} \land a^2-b^2 == 0$

**1:** $\int \frac{\cos[e+f x]^p \, (d \sin[e+f x])^n}{a+b \sin[e+f x]} \, dx$ when $\frac{p-1}{2} \in \mathbb{Z} \land a^2-b^2 == 0 \land n \in \mathbb{Z} \land \left(0<n<\frac{p+1}{2} \lor p \le -n < 2p-3 \lor 0<n \le -p\right)$

## Derivation: Algebraic expansion

Basis: If $a^2-b^2 == 0$, then $\frac{\cos[z]^2}{a+b \sin[z]} == \frac{1}{a} - \frac{d \sin[z]}{b d}$

Rule: If $\frac{p-1}{2} \in \mathbb{Z} \land a^2-b^2 == 0 \land n \in \mathbb{Z} \land \left(0<n<\frac{p+1}{2} \lor p \le -n < 2p-3 \lor 0<n \le -p\right)$, then

$$\int \frac{\cos[e+f x]^p \, (d \sin[e+f x])^n}{a+b \sin[e+f x]} \, dx \to \frac{1}{a} \int \cos[e+f x]^{p-2} \, (d \sin[e+f x])^n \, dx - \frac{1}{b d} \int \cos[e+f x]^{p-2} \, (d \sin[e+f x])^{n+1} \, dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(d_.*sin[e_.+f_.*x_])^n_./(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  1/a*Int[Cos[e+f*x]^(p-2)*(d*Sin[e+f*x])^n,x] -
  1/(b*d)*Int[Cos[e+f*x]^(p-2)*(d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,n,p},x] && IntegerQ[(p-1)/2] && EqQ[a^2-b^2,0] && IntegerQ[n] && (LtQ[0,n,(p+1)/2] || LeQ[p,-n] && LtQ[-n,2*p-3] ||
```

**2:** $\int \text{Cos}[e+f\,x]^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $\frac{p-1}{2}\in\mathbb{Z}\,\wedge\,a^2-b^2==0$

## Derivation: Integration by substitution

## Basis: If $\frac{p-1}{2}\in\mathbb{Z}\,\wedge\,a^2-b^2==0$, then

$$\text{Cos}[e+f\,x]^p\,(a+b\,\text{Sin}[e+f\,x])^m==\frac{1}{b^p\,f}\,\text{Subst}\Big[(a+x)^{m+\frac{p-1}{2}}\,(a-x)^{\frac{p-1}{2}},\,x,\,b\,\text{Sin}[e+f\,x]\Big]\,\partial_x\,(b\,\text{Sin}[e+f\,x])$$

## Rule: If $\frac{p-1}{2}\in\mathbb{Z}\,\wedge\,a^2-b^2==0$, then

$$\int \text{Cos}[e+f\,x]^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{1}{b^p\,f}\,\text{Subst}\Big[\int(a+x)^{m+\frac{p-1}{2}}\,(a-x)^{\frac{p-1}{2}}\left(c+\frac{d}{b}\,x\right)^n\,dx,\,x,\,b\,\text{Sin}[e+f\,x]\Big]$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  1/(b^p*f)*Subst[Int[(a+x)^(m+(p-1)/2)*(a-x)^((p-1)/2)*(c+d/b*x)^n,x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,c,d,m,n},x] && IntegerQ[(p-1)/2] && EqQ[a^2-b^2,0]
```

**4:** $\int \operatorname{Cos}[e+f\,x]^p\,(a+b\operatorname{Sin}[e+f\,x])^m\,(c+d\operatorname{Sin}[e+f\,x])^n\,dx$ when $\frac{p-1}{2}\in\mathbb{Z}\,\wedge\,a^2-b^2\neq 0$

## Derivation: Integration by substitution

Basis: If $\frac{p-1}{2}\in\mathbb{Z}$, then $\operatorname{Cos}[e+f\,x]^p == \frac{1}{b^p\,f}\operatorname{Subst}\left[\left(b^2-x^2\right)^{\frac{p-1}{2}},\,x,\,b\operatorname{Sin}[e+f\,x]\right]\partial_x\left(b\operatorname{Sin}[e+f\,x]\right)$

Rule: If $\frac{p-1}{2}\in\mathbb{Z}\,\wedge\,a^2-b^2\neq 0$, then

$$\int \operatorname{Cos}[e+f\,x]^p\,(a+b\operatorname{Sin}[e+f\,x])^m\,(c+d\operatorname{Sin}[e+f\,x])^n\,dx\,\rightarrow\,\frac{1}{b^p\,f}\operatorname{Subst}\left[\int (a+x)^m\left(c+\frac{d}{b}\,x\right)^n\left(b^2-x^2\right)^{\frac{p-1}{2}}dx,\,x,\,b\operatorname{Sin}[e+f\,x]\right]$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  1/(b^p*f)*Subst[Int[(a+x)^m*(c+d/b*x)^n*(b^2-x^2)^((p-1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IntegerQ[(p-1)/2] && NeQ[a^2-b^2,0]
```

**2:** $\int (g\operatorname{Cos}[e+f\,x])^p\,(d\operatorname{Sin}[e+f\,x])^n\,(a+b\operatorname{Sin}[e+f\,x])\,dx$

## Derivation: Algebraic expansion

## Rule:

$$\int (g\operatorname{Cos}[e+f\,x])^p\,(d\operatorname{Sin}[e+f\,x])^n\,(a+b\operatorname{Sin}[e+f\,x])\,dx\,\rightarrow\,a\int (g\operatorname{Cos}[e+f\,x])^p\,(d\operatorname{Sin}[e+f\,x])^n\,dx+\frac{b}{d}\int (g\operatorname{Cos}[e+f\,x])^p\,(d\operatorname{Sin}[e+f\,x])^{n+1}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_.*(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  a*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^n,x] + b/d*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,g,n,p},x]
```

**3:** $\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2==0$

Derivation: Algebraic expansion

Basis: If $a^2-b^2==0$, then $\frac{\operatorname{Cos}[z]^2}{a+b \operatorname{Sin}[z]}==\frac{1}{a}-\frac{d \operatorname{Sin}[z]}{b d}$

Rule: If $a^2-b^2==0$, then

$$\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx \longrightarrow \frac{g^2}{a} \int (g \operatorname{Cos}[e+f x])^{p-2} (d \operatorname{Sin}[e+f x])^n \, dx-\frac{g^2}{b d} \int (g \operatorname{Cos}[e+f x])^{p-2} (d \operatorname{Sin}[e+f x])^{n+1} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_./(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g^2/a*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^n,x] -
  g^2/(b*d)*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,g,n,p},x] && EqQ[a^2-b^2,0]
```

4. $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x])^n \, dx$ when $b c+a d == 0 \land a^2-b^2 == 0$

1: $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x])^n \, dx$ when $b c+a d == 0 \land a^2-b^2 == 0 \land m \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c+a d == 0 \land a^2-b^2 == 0$, then $(a+b \sin[z]) (c+d \sin[z]) == a c \cos[z]^2$

Rule: If $b c+a d == 0 \land a^2-b^2 == 0 \land m \in \mathbb{Z}$, then

$$\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x])^n \, dx \to \frac{a^m c^m}{g^{2 m}} \int (g \cos[e+f x])^{2 m+p} (c+d \sin[e+f x])^{n-m} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^m*c^m/g^(2*m)*Int[(g*Cos[e+f*x])^(2*m+p)*(c+d*Sin[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] && Not[IntegerQ[n] && LtQ[n^2,m^2]]
```

**2:** $\int \mathrm{Cos}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,\frac{p}{2}\in\mathbb{Z}$

## Derivation: Algebraic simplification

Basis: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then $\mathrm{Cos}[z]^2==\frac{1}{a\,c}\,(a+b\,\mathrm{Sin}[z])\,(c+d\,\mathrm{Sin}[z])$

Rule: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,\frac{p}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Cos}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{1}{a^{p/2}\,c^{p/2}}\int(a+b\,\mathrm{Sin}[e+f\,x])^{m+\frac{p}{2}}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+\frac{p}{2}}\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  1/(a^(p/2)*c^(p/2))*Int[(a+b*Sin[e+f*x])^(m+p/2)*(c+d*Sin[e+f*x])^(n+p/2),x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[p/2]
```

**3:** $\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0$

Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $\partial_x \frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}==0$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then

$$\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx \to \frac{g\,\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\int (g\,\mathrm{Cos}[e+f\,x])^{p-1}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_/(Sqrt[a_+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  g*Cos[e+f*x]/(Sqrt[a+b*Sin[e+f*x]]*Sqrt[c+d*Sin[e+f*x]])*Int[(g*Cos[e+f*x])^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

4. $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{p}{2}+\frac{1}{2}\in\mathbb{Z}^+$

   1. $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{p}{2}-\frac{1}{2}==0$

      **1:** $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,2\,m+p-1==0\,\land\,m-n-1==0$

Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $\partial_x \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^m}{(g\,\mathrm{Cos}[e+f\,x])^{2\,m}}==0$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,2\,m+p-1==0\,\land\,m-n-1==0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \to$$

$$\left(\left(a^{\text{IntPart}[m]} c^{\text{IntPart}[m]} (a+b \operatorname{Sin}[e+f x])^{\text{FracPart}[m]} (c+d \operatorname{Sin}[e+f x])^{\text{FracPart}[m]}\right) / \left(g^{2\,\text{IntPart}[m]} (g \operatorname{Cos}[e+f x])^{2\,\text{FracPart}[m]}\right)\right) \int \frac{(g \operatorname{Cos}[e+f x])^{2 m+p}}{c+d \operatorname{Sin}[e+f x]} \, dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^IntPart[m]*c^IntPart[m]*(a+b*Sin[e+f*x])^FracPart[m]*(c+d*Sin[e+f*x])^FracPart[m]/
    (g^(2*IntPart[m])*(g*Cos[e+f*x])^(2*FracPart[m]))*Int[(g*Cos[e+f*x])^(2*m+p)/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[2*m+p-1,0] && EqQ[m-n-1,0]
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $b c+a d=0 \land a^2-b^2=0 \land 2 m+p-1=0 \land m-n-1\neq 0$

## Derivation: Doubly degenerate sine recurrence 1a

## Rule: If $b c+a d=0 \land a^2-b^2=0 \land 2 m+p-1=0 \land m-n-1\neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \to \frac{b (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^{m-1} (c+d \operatorname{Sin}[e+f x])^n}{f g (m-n-1)}$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*g*(m-n-1)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[2*m+p-1,0] && NeQ[m-n-1,0]
```

2. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{p}{2}-\frac{1}{2}\in\mathbb{Z}^+$

1: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{p}{2}-\frac{1}{2}\in\mathbb{Z}^+\,\land\,n<-1$

## Derivation: Doubly degenerate sine recurrence 1a

## Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{p}{2}-\frac{1}{2}\in\mathbb{Z}^+\,\land\,n<-1$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{2\,b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n}{f\,g\,(2\,n+p+1)}-$$
$$\frac{b\,(2\,m+p-1)}{d\,(2\,n+p+1)}\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*g*(2*n+p+1)) -
  b*(2*m+p-1)/(d*(2*n+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[Simplify[m+p/2-1/2],0] && LtQ[n,-1] &&
  NeQ[2*n+p+1,0] && Not[ILtQ[Simplify[m+n+p],0] && GtQ[Simplify[2*m+n+3*p/2+1],0]]
```

2: $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{p}{2}-\frac{1}{2}\in\mathbb{Z}^+\land\,n\nless-1$

## Derivation: Doubly degenerate sine recurrence 1b

## Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{p}{2}-\frac{1}{2}\in\mathbb{Z}^+\land\,n\nless-1$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$
$$-\frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n}{f\,g\,(m+n+p)}+$$

$$\frac{a\,(2\,m+p-1)}{m+n+p}\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*g*(m+n+p)) +
  a*(2*m+p-1)/(m+n+p)*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[Simplify[m+p/2-1/2],0] && Not[LtQ[n,-1]] &&
  Not[IGtQ[Simplify[n+p/2-1/2],0] && GtQ[m-n,0]] && Not[ILtQ[Simplify[m+n+p],0] && GtQ[Simplify[2*m+n+3*p/2+1],0]]
```

5. $\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+n+p\in\mathbb{Z}^-$

   1. $\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+n+p+1==0$

      1: $\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+n+p+1==0\,\land\,m-n==0$

## Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $\partial_x\frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^m}{(g\,\mathrm{Cos}[e+f\,x])^{2\,m}}==0$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,2\,m+p+1==0$, then

$$\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$

$$\left(\left(a^{\mathrm{IntPart}[m]}\,c^{\mathrm{IntPart}[m]}\,(a+b\,\mathrm{Sin}[e+f\,x])^{\mathrm{FracPart}[m]}\,(c+d\,\mathrm{Sin}[e+f\,x])^{\mathrm{FracPart}[m]}\right)/\left(g^{2\,\mathrm{IntPart}[m]}\,(g\,\mathrm{Cos}[e+f\,x])^{2\,\mathrm{FracPart}[m]}\right)\right)\int(g\,\mathrm{Cos}[e+f\,x])^{2\,m+p}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  a^IntPart[m]*c^IntPart[m]*(a+b*Sin[e+f*x])^FracPart[m]*(c+d*Sin[e+f*x])^FracPart[m]/
    (g^(2*IntPart[m])*(g*Cos[e+f*x])^(2*FracPart[m]))*Int[(g*Cos[e+f*x])^(2*m+p),x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[2*m+p+1,0]
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $b c+a d=0 \wedge a^2-b^2=0 \wedge m+n+p+1=0 \wedge m-n \neq 0$

Derivation: Doubly degenerate sine recurrence 1c with $n \to -m-p-1$

Rule: If $b c+a d=0 \wedge a^2-b^2=0 \wedge m+n+p+1=0 \wedge m-n \neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \to \frac{b (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n}{a f g (m-n)}$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*g*(m-n)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && EqQ[m+n+p+1,0] && NeQ[m,n]
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $b c+a d == 0 \land a^2-b^2 == 0 \land m+n+p+1 \in \mathbb{Z}^- \land 2 m+p+1 \neq 0$

## Derivation: Doubly degenerate sine recurrence 1c

## Rule: If $b c+a d == 0 \land a^2-b^2 == 0 \land m+n+p+1 \in \mathbb{Z}^- \land 2 m+p+1 \neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \longrightarrow$$
$$\frac{b (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n}{a f g (2 m+p+1)}+\frac{m+n+p+1}{a (2 m+p+1)} \int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^{m+1} (c+d \operatorname{Sin}[e+f x])^n \, dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*g*(2*m+p+1)) +
  (m+n+p+1)/(a*(2*m+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && ILtQ[Simplify[m+n+p+1],0] && NeQ[2*m+p+1,0] &&
  (SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]])
```

6. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $b c+a d == 0 \land a^2-b^2 == 0 \land (2 m \mid 2 n \mid 2 p) \in \mathbb{Z}$

1. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $b c+a d == 0 \land a^2-b^2 == 0 \land m>0$

**1:** $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $b c+a d == 0 \land a^2-b^2 == 0 \land m>0 \land n<-1 \land 2 n+p+1 \neq 0$

## Derivation: Doubly degenerate sine recurrence 1a

## Rule: If $b c+a d == 0 \land a^2-b^2 == 0 \land m>0 \land n<-1 \land 2 n+p+1 \neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \longrightarrow$$
$$-\frac{2 b (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^{m-1} (c+d \operatorname{Sin}[e+f x])^n}{f g (2 n+p+1)}-$$

$$\frac{b\,(2m+p-1)}{d\,(2n+p+1)}\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*g*(2*n+p+1)) -
  b*(2*m+p-1)/(d*(2*n+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && GtQ[m,0] && LtQ[n,-1] && NeQ[2*n+p+1,0] && IntegersQ[2*m,2*n,2*p]
```

**2:** $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\;\land\;a^2-b^2==0\;\land\;m>0\;\land\;m+n+p\neq 0$

## Derivation: Doubly degenerate sine recurrence 1b

## Rule: If $b\,c+a\,d==0\;\land\;a^2-b^2==0\;\land\;m>0\;\land\;m+n+p\neq 0$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$-\frac{b\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n}{f\,g\,(m+n+p)}+\frac{a\,(2m+p-1)}{m+n+p}\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n/(f*g*(m+n+p)) +
  a*(2*m+p-1)/(m+n+p)*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && GtQ[m,0] && NeQ[m+n+p,0] && Not[LtQ[0,n,m]] && IntegersQ[2*m,2*n
```

**2:** $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m<-1\,\land\,2\,m+p+1\neq0$

## Derivation: Doubly degenerate sine recurrence 1c

## Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m<-1\,\land\,2\,m+p+1\neq0$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{b\,(g\,\mathrm{Cos}[e+f\,x])^{p+1}\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{a\,f\,g\,(2\,m+p+1)}+\frac{m+n+p+1}{a\,(2\,m+p+1)}\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(a*f*g*(2*m+p+1)) +
  (m+n+p+1)/(a*(2*m+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && LtQ[m,-1] && NeQ[2*m+p+1,0] && Not[LtQ[m,n,-1]] &&
  IntegersQ[2*m,2*n,2*p]
```

**7:** $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\land a^2-b^2==0\land m\notin\mathbb{Z}\land n\notin\mathbb{Z}$

## Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0\land a^2-b^2==0$, then $\partial_x\frac{(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^m}{(g\,\text{Cos}[e+f\,x])^{2\,m}}==0$

Rule: If $b\,c+a\,d==0\land a^2-b^2==0\land m\notin\mathbb{Z}\land n\notin\mathbb{Z}$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\to$$
$$\left(\left(a^{\text{IntPart}[m]}\,c^{\text{IntPart}[m]}\,(a+b\,\text{Sin}[e+f\,x])^{\text{FracPart}[m]}\,(c+d\,\text{Sin}[e+f\,x])^{\text{FracPart}[m]}\right)/\left(g^{2\,\text{IntPart}[m]}\,(g\,\text{Cos}[e+f\,x])^{2\,\text{FracPart}[m]}\right)\right)$$
$$\int (g\,\text{Cos}[e+f\,x])^{2\,m+p}\,(c+d\,\text{Sin}[e+f\,x])^{n-m}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^IntPart[m]*c^IntPart[m]*(a+b*Sin[e+f*x])^FracPart[m]*(c+d*Sin[e+f*x])^FracPart[m]/
    (g^(2*IntPart[m])*(g*Cos[e+f*x])^(2*FracPart[m]))*
    Int[(g*Cos[e+f*x])^(2*m+p)*(c+d*Sin[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && (FractionQ[m] || Not[FractionQ[n]])
```

5. $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x]) \, dx$

1. $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x]) \, dx$ when $a^2-b^2=0$

1: $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x]) \, dx$ when $a^2-b^2=0 \land a d m+b c (m+p+1)=0$

Derivation: Singly degenerate sine recurrence 2c with $c \to 1, d \to 0, n \to 0$

Note: If $a^2-b^2=0 \land a d m+b c (m+p+1)=0$, then $m+p+1 \neq 0$.

Rule: If $a^2-b^2=0 \land a d m+b c (m+p+1)=0$, then

$$\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x]) \, dx \longrightarrow -\frac{d (g \cos[e+f x])^{p+1} (a+b \sin[e+f x])^m}{f g (m+p+1)}$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(f*g*(m+p+1)) /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && EqQ[a*d*m+b*c*(m+p+1),0]
```

**2:** $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx$ when $a^2-b^2==0\,\land\,m>-1\,\land\,p<-1$

## Derivation: Singly degenerate sine recurrence 4a with $c\to 1,\ d\to 0$

## Rule: If $a^2-b^2==0\,\land\,m>-1\,\land\,p<-1$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{(b\,c+a\,d)\,(g\,\mathrm{Cos}[e+f\,x])^{p+1}\,(a+b\,\mathrm{Sin}[e+f\,x])^m}{a\,f\,g\,(p+1)}+\frac{b\,(a\,d\,m+b\,c\,(m+p+1))}{a\,g^2\,(p+1)}\int (g\,\mathrm{Cos}[e+f\,x])^{p+2}\,(a+b\,\mathrm{Sin}[e+f\,x])^{m-1}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -(b*c+a*d)*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*(p+1)) +
  b*(a*d*m+b*c*(m+p+1))/(a*g^2*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[m,-1] && LtQ[p,-1]
```

**3:** $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx$ when $a^2-b^2==0\,\wedge\,\frac{2m+p+1}{2}\in\mathbb{Z}^+\wedge\,m+p+1\neq 0$

Derivation: Singly degenerate sine recurrence 2c with $c\to 1,\ d\to 0,\ n\to 0$

Rule: If $a^2-b^2==0\,\wedge\,\frac{2m+p+1}{2}\in\mathbb{Z}^+\wedge\,m+p+1\neq 0$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{d\,(g\,\mathrm{Cos}[e+f\,x])^{p+1}\,(a+b\,\mathrm{Sin}[e+f\,x])^m}{f\,g\,(m+p+1)}+\frac{a\,d\,m+b\,c\,(m+p+1)}{b\,(m+p+1)}\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(f*g*(m+p+1)) +
  (a*d*m+b*c*(m+p+1))/(b*(m+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && IGtQ[Simplify[(2*m+p+1)/2],0] && NeQ[m+p+1,0]
```

4. $\int \text{Cos}[e+f x]^2 \, (a+b\,\text{Sin}[e+f x])^m \, (c+d\,\text{Sin}[e+f x]) \, dx$ when $a^2-b^2==0 \land m<0$

1: $\int \text{Cos}[e+f x]^2 \, (a+b\,\text{Sin}[e+f x])^m \, (c+d\,\text{Sin}[e+f x]) \, dx$ when $a^2-b^2==0 \land m<-\frac{3}{2}$

## Rule: If $a^2-b^2==0 \land m<-\frac{3}{2}$, then

$$\int \text{Cos}[e+f x]^2 \, (a+b\,\text{Sin}[e+f x])^m \, (c+d\,\text{Sin}[e+f x]) \, dx \longrightarrow$$

$$\frac{2\,(b\,c-a\,d)\,\text{Cos}[e+f x]\,(a+b\,\text{Sin}[e+f x])^{m+1}}{b^2\,f\,(2\,m+3)}+\frac{1}{b^3\,(2\,m+3)}\int (a+b\,\text{Sin}[e+f x])^{m+2}\,(b\,c+2\,a\,d\,(m+1)-b\,d\,(2\,m+3)\,\text{Sin}[e+f x])\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  2*(b*c-a*d)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b^2*f*(2*m+3)) +
  1/(b^3*(2*m+3))*Int[(a+b*Sin[e+f*x])^(m+2)*(b*c+2*a*d*(m+1)-b*d*(2*m+3)*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2-b^2,0] && LtQ[m,-3/2]
```

**2:** $\int \text{Cos}[e+f\,x]^2\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$ when $a^2-b^2==0\,\land\,-\frac{3}{2}\le m<0$

## Rule: If $a^2-b^2==0\,\land\,-\frac{3}{2}\le m<0$, then

$$\int \text{Cos}[e+f\,x]^2\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{d\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+2}}{b^2\,f\,(m+3)}-\frac{1}{b^2\,(m+3)}\int(a+b\,\text{Sin}[e+f\,x])^{m+1}\,(b\,d\,(m+2)-a\,c\,(m+3)+(b\,c\,(m+3)-a\,d\,(m+4))\,\text{Sin}[e+f\,x])\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  d*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+2)/(b^2*f*(m+3)) -
  1/(b^2*(m+3))*Int[(a+b*Sin[e+f*x])^(m+1)*(b*d*(m+2)-a*c*(m+3)+(b*c*(m+3)-a*d*(m+4))*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[a^2-b^2,0] && GeQ[m,-3/2] && LtQ[m,0]
```

**5:** $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx$ when $a^2-b^2=0 \land (m<-1 \lor m+p\in\mathbb{Z}^-) \land 2m+p+1\neq 0$

Derivation: Singly degenerate sine recurrence 2a with $c \to 1, \ d \to 0$

Derivation: Singly degenerate sine recurrence 2b with $c \to 1, \ d \to 0$

Rule: If $a^2-b^2=0 \land (m<-1 \lor m+p\in\mathbb{Z}^-) \land 2m+p+1\neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx \longrightarrow$$
$$\frac{(b c-a d) (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^m}{a f g (2 m+p+1)}+\frac{a d m+b c (m+p+1)}{a b (2 m+p+1)} \int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^{m+1} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (b*c-a*d)*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*(2*m+p+1)) +
  (a*d*m+b*c*(m+p+1))/(a*b*(2*m+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && (LtQ[m,-1] || ILtQ[Simplify[m+p],0]) && NeQ[2*m+p+1,0]
```

**6:** $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$ when $a^2-b^2==0\,\wedge\,m+p+1\neq 0$

Derivation: Singly degenerate sine recurrence 2c with $c\to 1,\ d\to 0,\ n\to 0$

Rule: If $a^2-b^2==0\,\wedge\,m+p+1\neq 0$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{d\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^m}{f\,g\,(m+p+1)}+\frac{a\,d\,m+b\,c\,(m+p+1)}{b\,(m+p+1)}\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(f*g*(m+p+1)) +
  (a*d*m+b*c*(m+p+1))/(b*(m+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && NeQ[m+p+1,0]
```

2. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$ when $a^2-b^2\neq 0$

1. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$ when $a^2-b^2\neq 0\,\wedge\,m>0$

**1:** $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$ when $a^2-b^2\neq 0\,\wedge\,m>0\,\wedge\,p<-1$

Derivation: Nondegenerate sine recurrence 3a with $c\to 1,\ d\to 0,\ C\to 0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m>0\,\wedge\,p<-1$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^m\,(d+c\,\text{Sin}[e+f\,x])}{f\,g\,(p+1)}+$$

$$\frac{1}{g^2\,(p+1)}\int(g\,\text{Cos}[e+f\,x])^{p+2}\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(a\,c\,(p+2)+b\,d\,m+b\,c\,(m+p+2)\,\text{Sin}[e+f\,x])\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m*(d+c*Sin[e+f*x])/(f*g*(p+1)) +
  1/(g^2*(p+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-1)*Simp[a*c*(p+2)+b*d*m+b*c*(m+p+2)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[a^2-b^2,0] && GtQ[m,0] && LtQ[p,-1] && IntegerQ[2*m] &&
  Not[EqQ[m,1] && NeQ[c^2-d^2,0] && SimplerQ[c+d*x,a+b*x]]
```

**2:** $\int(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx$ when $a^2-b^2\neq 0\land m>0\land p\not<-1$

## Derivation: Nondegenerate sine recurrence 1b with $c\to 0,\ d\to 1,\ A\to 0,\ B\to A,\ C\to B,\ n\to -1$

## Rule: If $a^2-b^2\neq 0\ \land\ m>0\ \land\ p\not<-1$, then

$$\int(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx\longrightarrow$$
$$-\frac{d\,(g\,\text{Cos}[e+f\,x])^{p+1}\,(a+b\,\text{Sin}[e+f\,x])^m}{f\,g\,(m+p+1)}+$$
$$\frac{1}{m+p+1}\int(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^{m-1}\,(a\,c\,(m+p+1)+b\,d\,m+(a\,d\,m+b\,c\,(m+p+1))\,\text{Sin}[e+f\,x])\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(f*g*(m+p+1)) +
  1/(m+p+1)*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m-1)*Simp[a*c*(m+p+1)+b*d*m+(a*d*m+b*c*(m+p+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[a^2-b^2,0] && GtQ[m,0] && Not[LtQ[p,-1]] && IntegerQ[2*m] &&
  Not[EqQ[m,1] && NeQ[c^2-d^2,0] && SimplerQ[c+d*x,a+b*x]]
```

2. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx$ when $a^2-b^2 \neq 0 \land m<-1$

1: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx$ when $a^2-b^2 \neq 0 \land m<-1 \land p>1 \land m+p+1 \neq 0$

Derivation: Nondegenerate sine recurrence 2a with $c \to 0,\ d \to 1,\ A \to 0,\ B \to A,\ C \to B,\ n \to -1$

Rule: If $a^2-b^2 \neq 0 \land m<-1 \land p>1 \land m+p+1 \neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx \longrightarrow$$

$$\left(\left(g (g \operatorname{Cos}[e+f x])^{p-1} (a+b \operatorname{Sin}[e+f x])^{m+1} (b c (m+p+1)-a d p+b d (m+1) \operatorname{Sin}[e+f x])\right) / \left(b^2 f (m+1) (m+p+1)\right)\right)+$$
$$\frac{g^2 (p-1)}{b^2 (m+1) (m+p+1)} \int (g \operatorname{Cos}[e+f x])^{p-2} (a+b \operatorname{Sin}[e+f x])^{m+1} (b d (m+1)+(b c (m+p+1)-a d p) \operatorname{Sin}[e+f x]) \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)*(a+b*Sin[e+f*x])^(m+1)*(b*c*(m+p+1)-a*d*p+b*d*(m+1)*Sin[e+f*x])/(b^2*f*(m+1)*(m+p+1)) +
  g^2*(p-1)/(b^2*(m+1)*(m+p+1))*
    Int[(g*Cos[e+f*x])^(p-2)*(a+b*Sin[e+f*x])^(m+1)*Simp[b*d*(m+1)+(b*c*(m+p+1)-a*d*p)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[p,1] && NeQ[m+p+1,0] && IntegerQ[2*m]
```

2: $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx$ when $a^2-b^2 \neq 0 \land m<-1$

Derivation: Nondegenerate sine recurrence 1a with $c \to 1,\ d \to 0,\ C \to 0$

Derivation: Nondegenerate sine recurrence 1c with $c \to 1,\ d \to 0,\ C \to 0$

Rule: If $a^2-b^2 \neq 0 \land m<-1$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx \longrightarrow$$

$$-\frac{(b\,c-a\,d)\,(g\,\mathrm{Cos}[e+f\,x])^{p+1}\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{f\,g\,(a^2-b^2)\,(m+1)}+$$
$$\frac{1}{(a^2-b^2)\,(m+1)}\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,((a\,c-b\,d)\,(m+1)-(b\,c-a\,d)\,(m+p+2)\,\mathrm{Sin}[e+f\,x])\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -(b*c-a*d)*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)/(f*g*(a^2-b^2)*(m+1)) +
  1/((a^2-b^2)*(m+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1)*Simp[(a*c-b*d)*(m+1)-(b*c-a*d)*(m+p+2)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && IntegerQ[2*m]
```

**3:** $\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx$ when $a^2-b^2\neq0\land p>1\land m+p\neq0\land m+p+1\neq0$

## Derivation: Nondegenerate sine recurrence 2b with $c\to0,\ d\to1,\ A\to0,\ B\to A,\ C\to B,\ n\to-1$

## Rule: If $a^2-b^2\neq0\land p>1\land m+p\neq0\land m+p+1\neq0$, then

$$\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])\,dx\longrightarrow$$
$$((g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(b\,c\,(m+p+1)-a\,d\,p+b\,d\,(m+p)\,\mathrm{Sin}[e+f\,x]))\,/\,(b^2\,f\,(m+p)\,(m+p+1)))+$$
$$\frac{g^2\,(p-1)}{b^2\,(m+p)\,(m+p+1)}\int(g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(b\,(a\,d\,m+b\,c\,(m+p+1))+(a\,b\,c\,(m+p+1)-d\,(a^2\,p-b^2\,(m+p)))\,\mathrm{Sin}[e+f\,x])\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)*(a+b*Sin[e+f*x])^(m+1)*(b*c*(m+p+1)-a*d*p+b*d*(m+p)*Sin[e+f*x])/(b^2*f*(m+p)*(m+p+1)) +
  g^2*(p-1)/(b^2*(m+p)*(m+p+1))*
    Int[(g*Cos[e+f*x])^(p-2)*(a+b*Sin[e+f*x])^m*Simp[b*(a*d*m+b*c*(m+p+1))+(a*b*c*(m+p+1)-d*(a^2*p-b^2*(m+p)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[a^2-b^2,0] && GtQ[p,1] && NeQ[m+p,0] && NeQ[m+p+1,0] && IntegerQ[2*m]
```

**4:** $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx$ when $a^2-b^2 \neq 0 \land p < -1$

## Derivation: Nondegenerate sine recurrence 3b with $c \to 1$, $d \to 0$, $C \to 0$

## Rule: If $a^2-b^2 \neq 0 \land p < -1$, then

$$\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x]) \, dx \longrightarrow$$

$$\left(\left((g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^{m+1} (b c-a d-(a c-b d) \operatorname{Sin}[e+f x])\right) / \left(f g \left(a^2-b^2\right) (p+1)\right)\right) +$$

$$\frac{1}{g^2 \left(a^2-b^2\right) (p+1)} \int (g \operatorname{Cos}[e+f x])^{p+2} (a+b \operatorname{Sin}[e+f x])^m \left(c \left(a^2 (p+2)-b^2 (m+p+2)\right)+a b d m+b (a c-b d) (m+p+3) \operatorname{Sin}[e+f x]\right) dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)*(b*c-a*d-(a*c-b*d)*Sin[e+f*x])/(f*g*(a^2-b^2)*(p+1)) +
  1/(g^2*(a^2-b^2)*(p+1))*
    Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^m*Simp[c*(a^2*(p+2)-b^2*(m+p+2))+a*b*d*m+b*(a*c-b*d)*(m+p+3)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[a^2-b^2,0] && LtQ[p,-1] && IntegerQ[2*m]
```

**5:** $\int \frac{(g \cos[e+f x])^p (c+d \sin[e+f x])}{a+b \sin[e+f x]} \, dx$ when $a^2-b^2 \neq 0$

## Derivation: Algebraic expansion

Basis: $\frac{c+d z}{a+b z} == \frac{d}{b} + \frac{b c-a d}{b (a+b z)}$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{(g \cos[e+f x])^p (c+d \sin[e+f x])}{a+b \sin[e+f x]} \, dx \rightarrow \frac{d}{b} \int (g \cos[e+f x])^p \, dx + \frac{b c-a d}{b} \int \frac{(g \cos[e+f x])^p}{a+b \sin[e+f x]} \, dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(c_.+d_.*sin[e_.+f_.*x_])/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  d/b*Int[(g*Cos[e+f*x])^p,x] + (b*c-a*d)/b*Int[(g*Cos[e+f*x])^p/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[a^2-b^2,0]
```

**6:** $\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x]) \, dx$ when $a^2-b^2 \neq 0 \wedge c^2-d^2 == 0$

## Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(g \cos[e+f x])^{p-1}}{(1+\sin[e+f x])^{\frac{p-1}{2}} (1-\sin[e+f x])^{\frac{p-1}{2}}} == 0$

Basis: $\cos[e+f x] == \frac{1}{f} \partial_x \sin[e+f x]$

Rule: If $a^2-b^2 \neq 0 \wedge c^2-d^2 == 0$, then

$$\int (g \cos[e+f x])^p (a+b \sin[e+f x])^m (c+d \sin[e+f x]) \, dx \rightarrow$$

$$\frac{c\,g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}}{(1+\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}}\int \mathrm{Cos}[e+f\,x]\,\left(1+\frac{d}{c}\,\mathrm{Sin}[e+f\,x]\right)^{\frac{p+1}{2}}\left(1-\frac{d}{c}\,\mathrm{Sin}[e+f\,x]\right)^{\frac{p-1}{2}}(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\;\longrightarrow$$

$$\frac{c\,g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}}{f\,(1+\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}}\,\mathrm{Subst}\left[\int\left(1+\frac{d}{c}\,x\right)^{\frac{p+1}{2}}\left(1-\frac{d}{c}\,x\right)^{\frac{p-1}{2}}(a+b\,x)^m\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_]),x_Symbol] :=
  c*g*(g*Cos[e+f*x])^(p-1)/(f*(1+Sin[e+f*x])^((p-1)/2)*(1-Sin[e+f*x])^((p-1)/2))*
    Subst[Int[(1+d/c*x)^((p+1)/2)*(1-d/c*x)^((p-1)/2)*(a+b*x)^m,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && NeQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

6. $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(d\,\mathrm{Sin}[e+f\,x])^n\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0$

1: $\int \mathrm{Cos}[e+f\,x]^p\,(d\,\mathrm{Sin}[e+f\,x])^n\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\wedge\,m\in\mathbb{Z}\,\wedge\,2\,m+p==0$

## Derivation: Algebraic simplification

Basis: If $a^2-b^2==0\,\wedge\,m\in\mathbb{Z}\,\wedge\,2\,m+p==0$, then $\mathrm{Cos}[z]^p\,(a+b\,\mathrm{Sin}[z])^m==\frac{a^{2\,m}}{(a-b\,\mathrm{Sin}[z])^m}$

Rule: If $a^2-b^2==0\,\wedge\,m\in\mathbb{Z}\,\wedge\,2\,m+p==0$, then

$$\int \mathrm{Cos}[e+f\,x]^p\,(d\,\mathrm{Sin}[e+f\,x])^n\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,dx\;\longrightarrow\;a^{2\,m}\int\frac{(d\,\mathrm{Sin}[e+f\,x])^n}{(a-b\,\mathrm{Sin}[e+f\,x])^m}\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^(2*m)*Int[(d*Sin[e+f*x])^n/(a-b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0] && IntegersQ[m,p] && EqQ[2*m+p,0]
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m=p$

Derivation: Nondegenerate sine recurrence 1b with $A \to a^2$, $B \to 2 a b$, $C \to b^2$, $m \to 0$

Rule: If $a^2-b^2=0 \land m=p$, then

$$\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$
$$-\frac{(g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^{m+1}}{2 b f g (m+1)}+\frac{a}{2 g^2} \int (g \operatorname{Cos}[e+f x])^{p+2} (a+b \operatorname{Sin}[e+f x])^{m-1} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*sin[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)/(2*b*f*g*(m+1)) +
  a/(2*g^2)*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && EqQ[m-p,0]
```

**3:** $\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \wedge m+p+1=0$

## Derivation: ???

## Rule: If $a^2-b^2=0 \wedge m+p+1=0$, then

$$\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$
$$\frac{b (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^m}{a f g m}-\frac{1}{g^2} \int (g \operatorname{Cos}[e+f x])^{p+2} (a+b \operatorname{Sin}[e+f x])^m \, dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*sin[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*m) -
  1/g^2*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && EqQ[m+p+1,0]
```

4. $\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2==0 \land m \in \mathbb{Z}$

**1:** $\int \operatorname{Cos}[e+f x]^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2==0 \land m \in \mathbb{Z} \land \frac{p}{2} \in \mathbb{Z} \land m+\frac{p}{2}>0$

## Derivation: Algebraic expansion

Basis: If $a^2-b^2==0 \land \frac{p}{2} \in \mathbb{Z}$, then $\operatorname{Cos}[z]^p == \frac{1}{a^p} (a-b \operatorname{Sin}[z])^{p/2} (a+b \operatorname{Sin}[z])^{p/2}$

Rule: If $a^2-b^2==0 \land m \in \mathbb{Z} \land \frac{p}{2} \in \mathbb{Z} \land m+\frac{p}{2}>0$, then

$$\int \operatorname{Cos}[e+f x]^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx \to$$
$$\frac{1}{a^p} \int \operatorname{ExpandTrig}\left[(d \operatorname{Sin}[e+f x])^n (a-b \operatorname{Sin}[e+f x])^{p/2} (a+b \operatorname{Sin}[e+f x])^{m+p/2}, x\right] dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  1/a^p*Int[ExpandTrig[(d*sin[e+f*x])^n*(a-b*sin[e+f*x])^(p/2)*(a+b*sin[e+f*x])^(m+p/2),x],x] /;
FreeQ[{a,b,d,e,f},x] && EqQ[a^2-b^2,0] && IntegersQ[m,n,p/2] && (GtQ[m,0] && GtQ[p,0] && LtQ[-m-p,n,-1] || GtQ[m,2] && LtQ[p,0] && G
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \wedge m \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

## Rule: If $a^2-b^2=0 \wedge m \in \mathbb{Z}^+$, then

$$\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx \to \int (g \operatorname{Cos}[e+f x])^p \operatorname{ExpandTrig}\left[(d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m, x\right] dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Int[ExpandTrig[(g*cos[e+f*x])^p,(d*sin[e+f*x])^n*(a+b*sin[e+f*x])^m,x],x] /;
FreeQ[{a,b,d,e,f,g,n,p},x] && EqQ[a^2-b^2,0] && IGtQ[m,0]
```

3. $\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \in \mathbb{Z}^-$

1: $\int \operatorname{Cos}[e+f x]^2 (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \in \mathbb{Z}^-$

## Derivation: Algebraic simplification

Basis: If $a^2-b^2=0$, then $\operatorname{Cos}[z]^2 = \frac{1}{b^2} (a+b \operatorname{Sin}[z]) (a-b \operatorname{Sin}[z])$

Rule: If $a^2-b^2=0 \land m \in \mathbb{Z}^-$, then

$$\int \operatorname{Cos}[e+f x]^2 (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx \rightarrow \frac{1}{b^2} \int (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^{m+1} (a-b \operatorname{Sin}[e+f x]) \, dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^2*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  1/b^2*Int[(d*Sin[e+f*x])^n*(a+b*Sin[e+f*x])^(m+1)*(a-b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && (ILtQ[m,0] || Not[IGtQ[n,0]])
```

**2:** $\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,m\in\mathbb{Z}^-$

## Derivation: Algebraic expansion

Basis: If $a^2-b^2==0$, then $a+b\,\text{Sin}[z]==\frac{a^2\,(g\,\text{Cos}[z])^2}{g^2\,(a-b\,\text{Sin}[z])}$

Rule: If $a^2-b^2==0\,\land\,m\in\mathbb{Z}^-$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow\,\frac{a^{2\,m}}{g^{2\,m}}\int\frac{(g\,\text{Cos}[e+f\,x])^{2\,m+p}\,(d\,\text{Sin}[e+f\,x])^n}{(a-b\,\text{Sin}[e+f\,x])^m}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (a/g)^(2*m)*Int[(g*Cos[e+f*x])^(2*m+p)*(d*Sin[e+f*x])^n/(a-b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,g,n,p},x] && EqQ[a^2-b^2,0] && ILtQ[m,0]
```

5: $\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2=0\,\land\,m\in\mathbb{Z}\,\land\,(2\,m+p=0\,\lor\,2\,m+p>0\,\land\,p<-1)$

Derivation: Algebraic expansion

Basis: If $a^2-b^2=0$, then $a+b\,\text{Sin}[z]=\frac{a^2\,(g\,\text{Cos}[z])^2}{g^2\,(a-b\,\text{Sin}[z])}$

Note: By making the degree of the cosine factor in the integrand nonnegative, this rule removes the removable singularities from the integrand and hence from the resulting antiderivatives.

Rule: If $a^2-b^2=0\,\land\,m\in\mathbb{Z}\,\land\,(2\,m+p=0\,\lor\,2\,m+p>0\,\land\,p<-1)$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow\,\frac{a^{2\,m}}{g^{2\,m}}\int\frac{(g\,\text{Cos}[e+f\,x])^{2\,m+p}\,(d\,\text{Sin}[e+f\,x])^n}{(a-b\,\text{Sin}[e+f\,x])^m}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (a/g)^(2*m)*Int[(g*Cos[e+f*x])^(2*m+p)*(d*Sin[e+f*x])^n/(a-b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,g,n},x] && EqQ[a^2-b^2,0] && IntegerQ[m] && RationalQ[p] && (EqQ[2*m+p,0] || GtQ[2*m+p,0] && LtQ[p,-1])
```

6. $\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0$

1: $\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \land m \le -\frac{1}{2}$

## Derivation: ???

## Rule: If $a^2-b^2=0 \land m \le -\frac{1}{2}$, then

$$\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$

$$\frac{b (g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^m}{a f g (2 m+p+1)}-\frac{1}{a^2 (2 m+p+1)} \int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^{m+1} (a m-b (2 m+p+1) \operatorname{Sin}[e+f x]) \, dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*sin[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^m/(a*f*g*(2*m+p+1)) -
  1/(a^2*(2*m+p+1))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^(m+1)*(a*m-b*(2*m+p+1)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && LeQ[m,-1/2] && NeQ[2*m+p+1,0]
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2=0 \wedge m \nless -\frac{1}{2}$

Derivation: Nondegenerate sine recurrence 1b with $A \to a^2$, $B \to 2 a b$, $C \to b^2$, $m \to 0$

Rule: If $a^2-b^2=0 \wedge m \nless -\frac{1}{2}$, then

$$\int (g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^2 (a+b \operatorname{Sin}[e+f x])^m \, dx \longrightarrow$$
$$-\frac{(g \operatorname{Cos}[e+f x])^{p+1} (a+b \operatorname{Sin}[e+f x])^{m+1}}{b f g (m+p+2)}+\frac{1}{b (m+p+2)} \int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (b (m+1)-a (p+1) \operatorname{Sin}[e+f x]) \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*sin[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m+1)/(b*f*g*(m+p+2)) +
  1/(b*(m+p+2))*Int[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^m*(b*(m+1)-a*(p+1)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && NeQ[m+p+2,0]
```

7. $\int \cos[e+f x]^p (d \sin[e+f x])^n (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2 == 0 \land \frac{p}{2} \in \mathbb{Z}$

1: $\int \cos[e+f x]^2 (d \sin[e+f x])^n (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2 == 0 \land (2m \mid 2n) \in \mathbb{Z}$

## Derivation: Algebraic simplification

Basis: If $a^2-b^2 == 0$, then $\cos[z]^2 == \frac{1}{b^2} (a+b \sin[z]) (a-b \sin[z])$

Rule: If $a^2-b^2 == 0 \land (2m \mid 2n) \in \mathbb{Z}$, then

$$\int \cos[e+f x]^2 (a+b \sin[e+f x])^m (d \sin[e+f x])^n \, dx \rightarrow \frac{1}{b^2} \int (d \sin[e+f x])^n (a+b \sin[e+f x])^{m+1} (a-b \sin[e+f x]) \, dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^2*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  1/b^2*Int[(d*Sin[e+f*x])^n*(a+b*Sin[e+f*x])^(m+1)*(a-b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && IntegersQ[2*m,2*n]
```

2. $\int \cos[e+fx]^4\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx$ when $a^2-b^2==0$

1: $\int \cos[e+fx]^4\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx$ when $a^2-b^2==0\ \wedge\ m<-1$

Derivation: Algebraic expansion

Basis: If $a^2-b^2==0$, then $\cos[z]^4 == -\frac{2}{a\,b}\sin[z]\,(a+b\sin[z])^2+\frac{1}{a^2}\,(1+\sin[z]^2)\,(a+b\sin[z])^2$

Rule: If $a^2-b^2==0\ \wedge\ 2m\in\mathbb{Z}\ \wedge\ m<-1$, then

$$\int \cos[e+fx]^4\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx\ \longrightarrow$$
$$-\frac{2}{a\,b\,d}\int (d\sin[e+fx])^{n+1}\,(a+b\sin[e+fx])^{m+2}\,dx+\frac{1}{a^2}\int (d\sin[e+fx])^n\,(a+b\sin[e+fx])^{m+2}\,(1+\sin[e+fx]^2)\,dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -2/(a*b*d)*Int[(d*Sin[e+f*x])^(n+1)*(a+b*Sin[e+f*x])^(m+2),x] +
  1/a^2*Int[(d*Sin[e+f*x])^n*(a+b*Sin[e+f*x])^(m+2)*(1+Sin[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0] && LtQ[m,-1]
```

2: $\int \cos[e+fx]^4\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx$ when $a^2-b^2==0\ \wedge\ m\nless -1$

Derivation: Algebraic expansion

Basis: $\cos[z]^4 == \sin[z]^4+1-2\sin[z]^2$

Rule: If $a^2-b^2==0\ \wedge\ m\nless -1$, then

$$\int \cos[e+fx]^4\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx\ \longrightarrow$$

$$\frac{1}{d^4}\int(d\,\text{Sin}[e+f\,x])^{n+4}\,(a+b\,\text{Sin}[e+f\,x])^m\,dx+\int(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,\left(1-2\,\text{Sin}[e+f\,x]^2\right)\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  1/d^4*Int[(d*Sin[e+f*x])^(n+4)*(a+b*Sin[e+f*x])^m,x] +
  Int[(d*Sin[e+f*x])^n*(a+b*Sin[e+f*x])^m*(1-2*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && Not[IGtQ[m,0]]
```

**3:** $\int\text{Cos}[e+f\,x]^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\wedge\,\frac{p}{2}\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$

## Derivation: Algebraic expansion, piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0\,\wedge\,\frac{p}{2}\in\mathbb{Z}$, then $\text{Cos}[z]^p==a^{-p}\,(a+b\,\text{Sin}[z])^{p/2}\,(a-b\,\text{Sin}[z])^{p/2}$

Basis: $\partial_x\frac{\text{Cos}[e+f\,x]}{\sqrt{1+\text{Sin}[e+f\,x]}\,\sqrt{1-\text{Sin}[e+f\,x]}}==0$

Basis: $\text{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\,\text{Sin}[e+f\,x]$

Rule: If $a^2-b^2==0\,\wedge\,\frac{p}{2}\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$, then

$$\int\text{Cos}[e+f\,x]^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$

$$a^{-p}\int(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^{m+p/2}\,(a-b\,\text{Sin}[e+f\,x])^{p/2}\,dx\,\longrightarrow$$

$$\frac{a^m\,\text{Cos}[e+f\,x]}{\sqrt{1+\text{Sin}[e+f\,x]}\,\sqrt{1-\text{Sin}[e+f\,x]}}\int\text{Cos}[e+f\,x]\,(d\,\text{Sin}[e+f\,x])^n\,\left(1+\frac{b}{a}\,\text{Sin}[e+f\,x]\right)^{m+\frac{p-1}{2}}\,\left(1-\frac{b}{a}\,\text{Sin}[e+f\,x]\right)^{\frac{p-1}{2}}\,dx\,\longrightarrow$$

$$\frac{a^m \operatorname{Cos}[e+f x]}{f \sqrt{1+\operatorname{Sin}[e+f x]} \sqrt{1-\operatorname{Sin}[e+f x]}} \operatorname{Subst}\left[\int (d x)^n \left(1+\frac{b}{a} x\right)^{m+\frac{p-1}{2}} \left(1-\frac{b}{a} x\right)^{\frac{p-1}{2}} dx, x, \operatorname{Sin}[e+f x]\right]$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  a^m*Cos[e+f*x]/(f*Sqrt[1+Sin[e+f*x]]*Sqrt[1-Sin[e+f*x]])*
    Subst[Int[(d*x)^n*(1+b/a*x)^(m+(p-1)/2)*(1-b/a*x)^((p-1)/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,d,e,f,n},x] && EqQ[a^2-b^2,0] && IntegerQ[p/2] && IntegerQ[m]
```

**4:** $\int \operatorname{Cos}[e+f x]^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m dx$ when $a^2-b^2=0 \land \frac{p}{2} \in \mathbb{Z} \land m \notin \mathbb{Z}$

## Derivation: Algebraic expansion, piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0 \land \frac{p}{2} \in \mathbb{Z}$, then $\operatorname{Cos}[z]^p = a^{-p} (a+b \operatorname{Sin}[z])^{p/2} (a-b \operatorname{Sin}[z])^{p/2}$

Basis: If $a^2-b^2=0$, then $\partial_x \frac{\operatorname{Cos}[e+f x]}{\sqrt{a+b \operatorname{Sin}[e+f x]} \sqrt{a-b \operatorname{Sin}[e+f x]}} = 0$

Basis: $\operatorname{Cos}[e+f x] = \frac{1}{f} \partial_x \operatorname{Sin}[e+f x]$

Rule: If $a^2-b^2=0 \land \frac{p}{2} \in \mathbb{Z} \land m \notin \mathbb{Z}$, then

$$\int \operatorname{Cos}[e+f x]^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m dx \longrightarrow$$

$$a^{-p} \int (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^{m+p/2} (a-b \operatorname{Sin}[e+f x])^{p/2} dx \longrightarrow$$

$$\frac{\operatorname{Cos}[e+f x]}{a^{p-2} \sqrt{a+b \operatorname{Sin}[e+f x]} \sqrt{a-b \operatorname{Sin}[e+f x]}} \int \operatorname{Cos}[e+f x] (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^{m+\frac{p}{2}-\frac{1}{2}} (a-b \operatorname{Sin}[e+f x])^{\frac{p}{2}-\frac{1}{2}} dx \longrightarrow$$

$$\frac{\text{Cos}[e+f x]}{a^{p-2} f \sqrt{a+b\,\text{Sin}[e+f x]}\ \sqrt{a-b\,\text{Sin}[e+f x]}}\ \text{Subst}\Big[\int (d\,x)^n\,(a+b\,x)^{m+\frac{p}{2}-\frac{1}{2}}\,(a-b\,x)^{\frac{p}{2}-\frac{1}{2}}\,dx,\ x,\ \text{Sin}[e+f x]\Big]$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Cos[e+f*x]/(a^(p-2)*f*Sqrt[a+b*Sin[e+f*x]]*Sqrt[a-b*Sin[e+f*x]])*
    Subst[Int[(d*x)^n(a+b*x)^(m+p/2-1/2)*(a-b*x)^(p/2-1/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && IntegerQ[p/2] && Not[IntegerQ[m]]
```

8: $\int (g\,\text{Cos}[e+f x])^p\,(d\,\text{Sin}[e+f x])^n\,(a+b\,\text{Sin}[e+f x])^m\,dx$ when $a^2-b^2==0\ \land\ m\in\mathbb{Z}^+$

## Derivation: Algebraic expansion

## Rule: If $a^2-b^2==0\ \land\ m\in\mathbb{Z}^+$, then

$$\int (g\,\text{Cos}[e+f x])^p\,(d\,\text{Sin}[e+f x])^n\,(a+b\,\text{Sin}[e+f x])^m\,dx\ \longrightarrow\ \int (g\,\text{Cos}[e+f x])^p\,\text{ExpandTrig}\big[(d\,\text{Sin}[e+f x])^n\,(a+b\,\text{Sin}[e+f x])^m,\ x\big]\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Int[ExpandTrig[(g*cos[e+f*x])^p,(d*sin[e+f*x])^n*(a+b*sin[e+f*x])^m,x],x] /;
FreeQ[{a,b,d,e,f,g,n,p},x] && EqQ[a^2-b^2,0] && IGtQ[m,0] && (IntegerQ[p] || IGtQ[n,0])
```

9. $\int (g\,\text{Cos}[e+f x])^p\,(d\,\text{Sin}[e+f x])^n\,(a+b\,\text{Sin}[e+f x])^m\,dx$ when $a^2-b^2==0$

1: $\int (g\,\text{Cos}[e+f x])^p\,(d\,\text{Sin}[e+f x])^n\,(a+b\,\text{Sin}[e+f x])^m\,dx$ when $a^2-b^2==0\ \land\ m\in\mathbb{Z}$

## Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(g\,\text{Cos}[e+f\,x])^{p-1}}{(1+\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\text{Sin}[e+f\,x])^{\frac{p-1}{2}}} == 0$

Basis: $\text{Cos}[e+f\,x] == \frac{1}{f}\,\partial_x\,\text{Sin}[e+f\,x]$

Rule: If $a^2-b^2==0\,\wedge\,m\in\mathbb{Z}$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$

$$\frac{a^m\,g\,(g\,\text{Cos}[e+f\,x])^{p-1}}{(1+\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\text{Sin}[e+f\,x])^{\frac{p-1}{2}}}\int \text{Cos}[e+f\,x]\,(d\,\text{Sin}[e+f\,x])^n\left(1+\frac{b}{a}\,\text{Sin}[e+f\,x]\right)^{m+\frac{p-1}{2}}\left(1-\frac{b}{a}\,\text{Sin}[e+f\,x]\right)^{\frac{p-1}{2}}dx\,\longrightarrow$$

$$\frac{a^m\,g\,(g\,\text{Cos}[e+f\,x])^{p-1}}{f\,(1+\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\text{Sin}[e+f\,x])^{\frac{p-1}{2}}}\,\text{Subst}\left[\int (d\,x)^n\left(1+\frac{b}{a}\,x\right)^{m+\frac{p-1}{2}}\left(1-\frac{b}{a}\,x\right)^{\frac{p-1}{2}}dx,\,x,\,\text{Sin}[e+f\,x]\right]$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  a^m*g*(g*Cos[e+f*x])^(p-1)/(f*(1+Sin[e+f*x])^((p-1)/2)*(1-Sin[e+f*x])^((p-1)/2))*
    Subst[Int[(d*x)^n*(1+b/a*x)^(m+(p-1)/2)*(1-b/a*x)^((p-1)/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,d,e,f,n,p},x] && EqQ[a^2-b^2,0] && IntegerQ[m]
```

**2:** $\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2==0\,\land\,m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0$, then $\partial_x \frac{\text{Cos}[e+f\,x]}{\sqrt{a+b\,\text{Sin}[e+f\,x]}\,\sqrt{a-b\,\text{Sin}[e+f\,x]}}==0$

Basis: $\text{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\,\text{Sin}[e+f\,x]$

Rule: If $a^2-b^2==0\,\land\,m\notin\mathbb{Z}$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx\,\longrightarrow$$

$$\frac{g\,(g\,\text{Cos}[e+f\,x])^{p-1}}{(a+b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}}\int \text{Cos}[e+f\,x]\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^{m+\frac{p-1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,dx\,\longrightarrow$$

$$\frac{g\,(g\,\text{Cos}[e+f\,x])^{p-1}}{f\,(a+b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(a-b\,\text{Sin}[e+f\,x])^{\frac{p-1}{2}}}\,\text{Subst}\Big[\int (d\,x)^n\,(a+b\,x)^{m+\frac{p-1}{2}}\,(a-b\,x)^{\frac{p-1}{2}}\,dx,\,x,\,\text{Sin}[e+f\,x]\Big]$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)/(f*(a+b*Sin[e+f*x])^((p-1)/2)*(a-b*Sin[e+f*x])^((p-1)/2))*
    Subst[Int[(d*x)^n*(a+b*x)^(m+(p-1)/2)*(a-b*x)^((p-1)/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,d,e,f,m,n,p},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]]
```

7. $\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a+b\,\text{Sin}[e+f\,x])^m\,dx$ when $a^2-b^2\neq 0$

1. $\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m}{\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

1: $\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m}{\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,m+p+\frac{1}{2}==0$

Rule: If $a^2-b^2\neq 0\,\wedge\,m<-1\,\wedge\,m+p+\frac{1}{2}==0$, then

$$\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m}{\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx\,\longrightarrow$$

$$-\frac{g\,(g\,\text{Cos}[e+f\,x])^{p-1}\,\sqrt{d\,\text{Sin}[e+f\,x]}\,\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{a\,d\,f\,(m+1)}+\frac{g^2\,(2\,m+3)}{2\,a\,(m+1)}\int \frac{(g\,\text{Cos}[e+f\,x])^{p-2}\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_/Sqrt[d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -g*(g*Cos[e+f*x])^(p-1)*Sqrt[d*Sin[e+f*x]]*(a+b*Sin[e+f*x])^(m+1)/(a*d*f*(m+1)) +
  g^2*(2*m+3)/(2*a*(m+1))*Int[(g*Cos[e+f*x])^(p-2)*(a+b*Sin[e+f*x])^(m+1)/Sqrt[d*Sin[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && EqQ[m+p+1/2,0]
```

**2:** $\int \frac{(g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m}{\sqrt{d \operatorname{Sin}[e+f x]}} \, dx$ when $a^2-b^2 \neq 0 \land m>0 \land m+p+\frac{3}{2}==0$

Rule: If $a^2-b^2 \neq 0 \land m>0 \land m+p+\frac{3}{2}==0$, then

$$\int \frac{(g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m}{\sqrt{d \operatorname{Sin}[e+f x]}} \, dx \longrightarrow$$

$$\frac{2 (g \operatorname{Cos}[e+f x])^{p+1} \sqrt{d \operatorname{Sin}[e+f x]} \, (a+b \operatorname{Sin}[e+f x])^m}{d f g (2 m+1)}+\frac{2 a m}{g^2 (2 m+1)} \int \frac{(g \operatorname{Cos}[e+f x])^{p+2} (a+b \operatorname{Sin}[e+f x])^{m-1}}{\sqrt{d \operatorname{Sin}[e+f x]}} \, dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_/Sqrt[d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  2*(g*Cos[e+f*x])^(p+1)*Sqrt[d*Sin[e+f*x]]*(a+b*Sin[e+f*x])^m/(d*f*g*(2*m+1)) +
  2*a*m/(g^2*(2*m+1))*Int[(g*Cos[e+f*x])^(p+2)*(a+b*Sin[e+f*x])^(m-1)/Sqrt[d*Sin[e+f*x]],x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && GtQ[m,0] && EqQ[m+p+3/2,0]
```

2. $\int \text{Cos}[e+f x]^p \, (d \, \text{Sin}[e+f x])^n \, (a+b \, \text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z}) \land \frac{p}{2} \in \mathbb{Z}^+$

1: $\int \text{Cos}[e+f x]^2 \, (d \, \text{Sin}[e+f x])^n \, (a+b \, \text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z})$

## Derivation: Algebraic expansion

Basis: $\text{Cos}[z]^2 = 1-\text{Sin}[z]^2$

Rule: If $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z})$, then

$$\int \text{Cos}[e+f x]^2 \, (d \, \text{Sin}[e+f x])^n \, (a+b \, \text{Sin}[e+f x])^m \, dx \to \int (d \, \text{Sin}[e+f x])^n \, (a+b \, \text{Sin}[e+f x])^m \left(1-\text{Sin}[e+f x]^2\right) dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^2*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Int[(d*Sin[e+f*x])^n*(a+b*Sin[e+f*x])^m*(1-Sin[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,m,n},x] && NeQ[a^2-b^2,0] && (IGtQ[m,0] || IntegersQ[2*m,2*n])
```

2. $\int \text{Cos}[e+f x]^4 \, (d \, \text{Sin}[e+f x])^n \, (a+b \, \text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z})$

1. $\int \text{Cos}[e+f x]^4 \, (d \, \text{Sin}[e+f x])^n \, (a+b \, \text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z}) \land m < -1$

x: $\int \text{Cos}[e+f x]^4 \, (d \, \text{Sin}[e+f x])^n \, (a+b \, \text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land m < -1 \land n < -1$

## Derivation: Algebraic expansion

Basis: $\text{Cos}[z]^4 = 1-2\,\text{Sin}[z]^2+\text{Sin}[z]^4$

Note: This produces a slightly simpler antiderivative when $m = -2$.

Rule: If $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land m < -1 \land n < -1$, then

$$\int \text{Cos}[e+f x]^4 (d\,\text{Sin}[e+f x])^n (a+b\,\text{Sin}[e+f x])^m\,dx \longrightarrow$$

$$\frac{(a^2-b^2)\,\text{Cos}[e+f x]\,\text{Sin}[e+f x]^{n+1} (a+b\,\text{Sin}[e+f x])^{m+1}}{a\,b^2\,d\,(m+1)} -$$

$$\left(\left(a^2 (n+1)-b^2 (m+n+2)\right) \text{Cos}[e+f x]\,\text{Sin}[e+f x]^{n+1} (a+b\,\text{Sin}[e+f x])^{m+2}\right) \Big/ \left(a^2 b^2 d\,(n+1)\,(m+1)\right) +$$

$$\frac{1}{a^2\,b\,(n+1)\,(m+1)} \int \text{Sin}[e+f x]^{n+1} (a+b\,\text{Sin}[e+f x])^{m+1} \cdot$$

$$\left(a^2 (n+1)(n+2)-b^2 (m+n+2)(m+n+3)+a\,b\,(m+1)\,\text{Sin}[e+f x]-\left(a^2 (n+1)(n+3)-b^2 (m+n+2)(m+n+4)\right) \text{Sin}[e+f x]^2\right) dx$$

## Program code:

```
(* Int[cos[e_.+f_.*x_]^4*sin[e_.+f_.*x_]^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (a^2-b^2)*Cos[e+f*x]*Sin[e+f*x]^(n+1)*(a+b*Sin[e+f*x])^(m+1)/(a*b^2*d*(m+1)) -
  (a^2*(n+1)-b^2*(m+n+2))*Cos[e+f*x]*Sin[e+f*x]^(n+1)*(a+b*Sin[e+f*x])^(m+2)/(a^2*b^2*d*(n+1)*(m+1)) +
  1/(a^2*b*(n+1)*(m+1))*Int[Sin[e+f*x]^(n+1)*(a+b*Sin[e+f*x])^(m+1)*
    Simp[a^2*(n+1)*(n+2)-b^2*(m+n+2)*(m+n+3)+a*b*(m+1)*Sin[e+f*x]-(a^2*(n+1)*(n+3)-b^2*(m+n+2)*(m+n+4))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && IntegersQ[2*m,2*n] && LtQ[m,-1] && LtQ[n,-1] *)
```

**1:** $\int \text{Cos}[e+f x]^4 (d\,\text{Sin}[e+f x])^n (a+b\,\text{Sin}[e+f x])^m\,dx$ when $a^2-b^2\neq 0 \land (2m \mid 2n)\in\mathbb{Z} \land m<-1 \land n<-1$

Derivation: Algebraic expansion and sine recurrence 3b with $A\to 1,\ B\to 0,\ C\to -2,\ m\to n,\ n\to p$, 2b with $A\to -b\,(m+n+2),\ B\to -a\,n,\ C\to b\,(n+p+3),\ m\to n+1,\ n\to p$ and 2a with $A\to 0,\ B\to 0,\ C\to 1,\ m\to n+4-2,\ n\to p$

Basis: $\cos[z]^4 = 1-2\,\text{Sin}[z]^2+\text{Sin}[z]^4$

Rule: If $a^2-b^2\neq 0 \land (2m \mid 2n)\in\mathbb{Z} \land m<-1 \land n<-1$, then

$$\int \text{Cos}[e+f x]^4 (d\,\text{Sin}[e+f x])^n (a+b\,\text{Sin}[e+f x])^m\,dx \longrightarrow$$

$$\int (d\,\text{Sin}[e+f x])^n (a+b\,\text{Sin}[e+f x])^m \left(1-2\,\text{Sin}[e+f x]^2\right) dx+\frac{1}{d^4}\int (d\,\text{Sin}[e+f x])^{n+4} (a+b\,\text{Sin}[e+f x])^m\,dx \longrightarrow$$

$$\frac{\text{Cos}[e+f\,x]\,(d\,\text{Sin}[e+f\,x])^{n+1}\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{a\,d\,f\,(n+1)}-$$
$$\left(\left(a^2\,(n+1)-b^2\,(m+n+2)\right)\text{Cos}[e+f\,x]\,(d\,\text{Sin}[e+f\,x])^{n+2}\,(a+b\,\text{Sin}[e+f\,x])^{m+1}\right)\Big/\left(a^2\,b\,d^2\,f\,(n+1)\,(m+1)\right)+$$
$$\frac{1}{a^2\,b\,d\,(n+1)\,(m+1)}\int(d\,\text{Sin}[e+f\,x])^{n+1}\,(a+b\,\text{Sin}[e+f\,x])^{m+1}\cdot$$
$$\left(\left(a^2\,(n+1)\,(n+2)-b^2\,(m+n+2)\,(m+n+3)+a\,b\,(m+1)\,\text{Sin}[e+f\,x]-\left(a^2\,(n+1)\,(n+3)-b^2\,(m+n+2)\,(m+n+4)\right)\text{Sin}[e+f\,x]^2\right)\right)dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Cos[e+f*x]*(d*Sin[e+f*x])^(n+1)*(a+b*Sin[e+f*x])^(m+1)/(a*d*f*(n+1)) -
  (a^2*(n+1)-b^2*(m+n+2))*Cos[e+f*x]*(d*Sin[e+f*x])^(n+2)*(a+b*Sin[e+f*x])^(m+1)/(a^2*b*d^2*f*(n+1)*(m+1)) +
  1/(a^2*b*d*(n+1)*(m+1))*Int[(d*Sin[e+f*x])^(n+1)*(a+b*Sin[e+f*x])^(m+1)*
    Simp[a^2*(n+1)*(n+2)-b^2*(m+n+2)*(m+n+3)+a*b*(m+1)*Sin[e+f*x]-(a^2*(n+1)*(n+3)-b^2*(m+n+2)*(m+n+4))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && IntegersQ[2*m,2*n] && LtQ[m,-1] && LtQ[n,-1]
```

2. $\int \text{Cos}[e+f x]^4 (d \text{Sin}[e+f x])^n (a+b \text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land m<-1 \land n \nless -1$

**1:** $\int \text{Cos}[e+f x]^4 (d \text{Sin}[e+f x])^n (a+b \text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land m<-1 \land n \nless -1 \land (m<-2 \lor m+n+4=0)$

Derivation: Algebraic expansion

Basis: $\text{Cos}[z]^4 = 1-2 \text{Sin}[z]^2+\text{Sin}[z]^4$

Rule: If $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land m<-1 \land n \nless -1 \land (m<-2 \lor m+n+4=0)$, then

$$\int \text{Cos}[e+f x]^4 (d \text{Sin}[e+f x])^n (a+b \text{Sin}[e+f x])^m \, dx \rightarrow$$

$$\frac{(a^2-b^2) \text{Cos}[e+f x] (d \text{Sin}[e+f x])^{n+1} (a+b \text{Sin}[e+f x])^{m+1}}{a b^2 d f (m+1)}+$$

$$\left((a^2 (n-m+1)-b^2 (m+n+2)) \text{Cos}[e+f x] (d \text{Sin}[e+f x])^{n+1} (a+b \text{Sin}[e+f x])^{m+2}\right) / (a^2 b^2 d f (m+1) (m+2))-$$

$$\frac{1}{a^2 b^2 (m+1) (m+2)} \int (d \text{Sin}[e+f x])^n (a+b \text{Sin}[e+f x])^{m+2} \cdot$$

$$\left(a^2 (n+1) (n+3)-b^2 (m+n+2) (m+n+3)+a b (m+2) \text{Sin}[e+f x]-(a^2 (n+2) (n+3)-b^2 (m+n+2) (m+n+4)) \text{Sin}[e+f x]^2\right) dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (a^2-b^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(d*Sin[e+f*x])^(n+1)/(a*b^2*d*f*(m+1)) +
  (a^2*(n-m+1)-b^2*(m+n+2))*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+2)*(d*Sin[e+f*x])^(n+1)/(a^2*b^2*d*f*(m+1)*(m+2)) -
  1/(a^2*b^2*(m+1)*(m+2))*Int[(a+b*Sin[e+f*x])^(m+2)*(d*Sin[e+f*x])^n*
    Simp[a^2*(n+1)*(n+3)-b^2*(m+n+2)*(m+n+3)+a*b*(m+2)*Sin[e+f*x]-(a^2*(n+2)*(n+3)-b^2*(m+n+2)*(m+n+4))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0] && IntegersQ[2*m,2*n] && LtQ[m,-1] && Not[LtQ[n,-1]] && (LtQ[m,-2] || EqQ[m+n+4,0])
```

**2:** $\int \text{Cos}[e+f x]^4 \, (d \,\text{Sin}[e+f x])^n \, (a+b \,\text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land m < -1 \land n \nless -1 \land m+n+4 \neq 0$

## Derivation: Algebraic expansion

## Basis: $\text{Cos}[z]^4 == 1 - 2\,\text{Sin}[z]^2 + \text{Sin}[z]^4$

## Rule: If $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land m < -1 \land n \nless -1 \land m+n+4 \neq 0$, then

$$\int \text{Cos}[e+f x]^4 \, (d \,\text{Sin}[e+f x])^n \, (a+b \,\text{Sin}[e+f x])^m \, dx \to$$

$$\frac{(a^2-b^2)\,\text{Cos}[e+f x]\,(d\,\text{Sin}[e+f x])^{n+1}\,(a+b\,\text{Sin}[e+f x])^{m+1}}{a\,b^2\,d\,f\,(m+1)} - \frac{\text{Cos}[e+f x]\,(d\,\text{Sin}[e+f x])^{n+1}\,(a+b\,\text{Sin}[e+f x])^{m+2}}{b^2\,d\,f\,(m+n+4)} -$$

$$\frac{1}{a\,b^2\,(m+1)\,(m+n+4)} \int (d\,\text{Sin}[e+f x])^n\,(a+b\,\text{Sin}[e+f x])^{m+1} \cdot$$

$$\left(a^2\,(n+1)\,(n+3) - b^2\,(m+n+2)\,(m+n+4) + a\,b\,(m+1)\,\text{Sin}[e+f x] - \left(a^2\,(n+2)\,(n+3) - b^2\,(m+n+3)\,(m+n+4)\right)\,\text{Sin}[e+f x]^2\right) dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  (a^2-b^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(d*Sin[e+f*x])^(n+1)/(a*b^2*d*f*(m+1)) -
  Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+2)*(d*Sin[e+f*x])^(n+1)/(b^2*d*f*(m+n+4)) -
  1/(a*b^2*(m+1)*(m+n+4))*Int[(a+b*Sin[e+f*x])^(m+1)*(d*Sin[e+f*x])^n*
    Simp[a^2*(n+1)*(n+3)-b^2*(m+n+2)*(m+n+4)+a*b*(m+1)*Sin[e+f*x]-(a^2*(n+2)*(n+3)-b^2*(m+n+3)*(m+n+4))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,n},x] && NeQ[a^2-b^2,0] && IntegersQ[2*m,2*n] && LtQ[m,-1] && Not[LtQ[n,-1]] && NeQ[m+n+4,0]
```

2. $\int \text{Cos}[e+f x]^4 \, (d \,\text{Sin}[e+f x])^n \, (a+b \,\text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z}) \land m \nless -1$

1. $\int \text{Cos}[e+f x]^4 \, (d \,\text{Sin}[e+f x])^n \, (a+b \,\text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z}) \land m \nless -1 \land n < -1$

**1:**

$\int \text{Cos}[e+f x]^4 \, (d \,\text{Sin}[e+f x])^n \, (a+b \,\text{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z}) \land m \nless -1 \land n < -1 \land (n < -2 \lor m+n+4 == 0)$

Derivation: Algebraic expansion and sine recurrence 3b with $A \to 1$, $B \to 0$, $C \to 0$, $m \to n$, $n \to p$ and 3b with $A \to -b\,(n+p+2)$, $B \to a\,(n+2)$, $C \to b\,(n+p+3)$, $m \to n+1$, $n \to p$

Basis: $\cos[z]^4 == 1 - 2\sin[z]^2 + \sin[z]^4$

Rule: If $a^2 - b^2 \neq 0 \wedge (m \in \mathbb{Z}^+ \vee (2m \mid 2n) \in \mathbb{Z}) \wedge m \not< -1 \wedge n < -1 \wedge (n < -2 \vee m + n + 4 == 0)$, then

$$\int \cos[e+fx]^4\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx \longrightarrow$$

$$\int (d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx - \frac{1}{d^2}\int (d\sin[e+fx])^{n+2}\,(a+b\sin[e+fx])^m\,(2-\sin[e+fx]^2)\,dx \longrightarrow$$

$$\frac{\cos[e+fx]\,(d\sin[e+fx])^{n+1}\,(a+b\sin[e+fx])^{m+1}}{a\,d\,f\,(n+1)} - \frac{b\,(m+n+2)\cos[e+fx]\,(d\sin[e+fx])^{n+2}\,(a+b\sin[e+fx])^{m+1}}{a^2\,d^2\,f\,(n+1)\,(n+2)} - \frac{1}{a^2\,d^2\,(n+1)\,(n+2)}\int (d\sin[e+fx])^{n+2}\,(a+b\sin[e+fx])^m \cdot$$
$$\left(a^2\,n\,(n+2) - b^2\,(m+n+2)\,(m+n+3) + a\,b\,m\sin[e+fx] - \left(a^2\,(n+1)\,(n+2) - b^2\,(m+n+2)\,(m+n+4)\right)\sin[e+fx]^2\right)dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(d*Sin[e+f*x])^(n+1)/(a*d*f*(n+1)) -
  b*(m+n+2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(d*Sin[e+f*x])^(n+2)/(a^2*d^2*f*(n+1)*(n+2)) -
  1/(a^2*d^2*(n+1)*(n+2))*Int[(a+b*Sin[e+f*x])^m*(d*Sin[e+f*x])^(n+2)*
    Simp[a^2*n*(n+2)-b^2*(m+n+2)*(m+n+3)+a*b*m*Sin[e+f*x]-(a^2*(n+1)*(n+2)-b^2*(m+n+2)*(m+n+4))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,m},x] && NeQ[a^2-b^2,0] && (IGtQ[m,0] || IntegersQ[2*m,2*n]) && Not[m<-1] && LtQ[n,-1] && (LtQ[n,-2] || EqQ[m+n+4,0
```

**2:** $\int \cos[e+fx]^4 (d\sin[e+fx])^n (a+b\sin[e+fx])^m \, dx$ when $a^2-b^2\neq 0 \wedge (m\in\mathbb{Z}^+ \vee (2m \mid 2n)\in\mathbb{Z}) \wedge m \nless -1 \wedge n<-1 \wedge m+n+4\neq 0$

Derivation: Algebraic expansion and sine recurrence 3b with $A\to 1,\ B\to 0,\ C\to -2,\ m\to n,\ n\to p$ and 3a with $A\to 0,\ B\to 0,\ C\to 1,\ m\to n+4-2,\ n\to p$

Basis: $\cos[z]^4 == 1-2\sin[z]^2+\sin[z]^4$

Rule: If $a^2-b^2\neq 0 \wedge (m\in\mathbb{Z}^+ \vee (2m \mid 2n)\in\mathbb{Z}) \wedge m \nless -1 \wedge n<-1 \wedge m+n+4\neq 0$, then

$$\int \cos[e+fx]^4 (d\sin[e+fx])^n (a+b\sin[e+fx])^m \, dx \longrightarrow$$

$$\int (d\sin[e+fx])^n (a+b\sin[e+fx])^m \left(1-2\sin[e+fx]^2\right) dx+\int (d\sin[e+fx])^{n+4} (a+b\sin[e+fx])^m \, dx \longrightarrow$$

$$\frac{\cos[e+fx]\,(d\sin[e+fx])^{n+1}(a+b\sin[e+fx])^{m+1}}{a\,d\,f\,(n+1)}-\frac{\cos[e+fx]\,(d\sin[e+fx])^{n+2}(a+b\sin[e+fx])^{m+1}}{b\,d^2\,f\,(m+n+4)}+$$
$$\frac{1}{a\,b\,d\,(n+1)\,(m+n+4)}\int (d\sin[e+fx])^{n+1}(a+b\sin[e+fx])^m\cdot$$
$$\left(a^2(n+1)(n+2)-b^2(m+n+2)(m+n+4)+a\,b\,(m+3)\sin[e+fx]-\left(a^2(n+1)(n+3)-b^2(m+n+3)(m+n+4)\right)\sin[e+fx]^2\right)dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(d*Sin[e+f*x])^(n+1)/(a*d*f*(n+1)) -
  Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(d*Sin[e+f*x])^(n+2)/(b*d^2*f*(m+n+4)) +
  1/(a*b*d*(n+1)*(m+n+4))*Int[(a+b*Sin[e+f*x])^m*(d*Sin[e+f*x])^(n+1)*
    Simp[a^2*(n+1)*(n+2)-b^2*(m+n+2)*(m+n+4)+a*b*(m+3)*Sin[e+f*x]-(a^2*(n+1)*(n+3)-b^2*(m+n+3)*(m+n+4))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,m},x] && NeQ[a^2-b^2,0] && (IGtQ[m,0] || IntegersQ[2*m,2*n]) && Not[m<-1] && LtQ[n,-1] && NeQ[m+n+4,0]
```

**2:**

$\int \cos[e+f x]^4 (d \sin[e+f x])^n (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z}) \land m \nless -1 \land n \nless -1 \land m+n+3 \neq 0 \land m+n+4 \neq 0$

Derivation: Algebraic expansion and sine recurrence 3a with $A \to 0,\ B \to 0,\ C \to 1,\ m \to n+4-2,\ n \to p$ and 3a with $A \to a\,(n+2),\ B \to b\,(n+p+3),\ C \to -a\,(n+3),\ m \to n+1,\ n \to p$

Basis: $\cos[z]^4 == 1-2 \sin[z]^2+\sin[z]^4$

Rule: If $a^2-b^2 \neq 0 \land (m \in \mathbb{Z}^+ \lor (2m \mid 2n) \in \mathbb{Z}) \land m \nless -1 \land n \nless -1 \land m+n+3 \neq 0 \land m+n+4 \neq 0$, then

$$\int \cos[e+f x]^4 (d \sin[e+f x])^n (a+b \sin[e+f x])^m \, dx \to$$

$$\int (d \sin[e+f x])^n (a+b \sin[e+f x])^m \left(1-2 \sin[e+f x]^2\right) dx + \frac{1}{d^4} \int (d \sin[e+f x])^{n+4} (a+b \sin[e+f x])^m \, dx \to$$

$$\frac{a\,(n+3) \cos[e+f x]\,(d \sin[e+f x])^{n+1} (a+b \sin[e+f x])^{m+1}}{b^2 d f\,(m+n+3)\,(m+n+4)} - \frac{\cos[e+f x]\,(d \sin[e+f x])^{n+2} (a+b \sin[e+f x])^{m+1}}{b\, d^2 f\,(m+n+4)} - \frac{1}{b^2\,(m+n+3)\,(m+n+4)} \int (d \sin[e+f x])^n (a+b \sin[e+f x])^m \cdot \left(a^2\,(n+1)\,(n+3) - b^2\,(m+n+3)\,(m+n+4) + a\, b\, m \sin[e+f x] - \left(a^2\,(n+2)\,(n+3) - b^2\,(m+n+3)\,(m+n+5)\right) \sin[e+f x]^2\right) dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^4*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  a*(n+3)*Cos[e+f*x]*(d*Sin[e+f*x])^(n+1)*(a+b*Sin[e+f*x])^(m+1)/(b^2*d*f*(m+n+3)*(m+n+4)) -
  Cos[e+f*x]*(d*Sin[e+f*x])^(n+2)*(a+b*Sin[e+f*x])^(m+1)/(b*d^2*f*(m+n+4)) -
  1/(b^2*(m+n+3)*(m+n+4))*Int[(d*Sin[e+f*x])^n*(a+b*Sin[e+f*x])^m*
    Simp[a^2*(n+1)*(n+3)-b^2*(m+n+3)*(m+n+4)+a*b*m*Sin[e+f*x]-(a^2*(n+2)*(n+3)-b^2*(m+n+3)*(m+n+5))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,m,n},x] && NeQ[a^2-b^2,0] && (IGtQ[m,0] || IntegersQ[2*m,2*n]) && Not[m<-1] && Not[LtQ[n,-1]] && NeQ[m+n+3,0] && Ne
```

**3:** $\int \cos[e+f x]^6 (d \sin[e+f x])^n (a+b \sin[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land n \neq -1 \land n \neq -2 \land m+n+5 \neq 0 \land m+n+6 \neq 0$

Derivation: Algebraic expansion and sine recurrence 3b with $A \to 1,\ B \to 0,\ C \to -3,\ m \to n,\ n \to p$, 3b with $A \to -b\,(2+n+p),\ B \to a\,(2+n-3\,(1+n)),\ C \to b\,(3+n+p),\ m \to n+1,\ n \to p$, 3a with $A \to 3,\ B \to 0,\ C \to -1,\ m \to n+4,\ n \to p$ and 3a with $A \to -a\,(4+n),\ B \to b\,(-5-n-p+3\,(6+n+p)),\ C \to a\,(5+n),\ m \to n+3,\ n \to p$

Basis: $\cos[z]^6 == 1 - 3\sin[z]^2 + \sin[z]^4\,(3 - \sin[z]^2)$

Rule: If $a^2 - b^2 \neq 0 \land (2m \mid 2n) \in \mathbb{Z} \land n \neq -1 \land n \neq -2 \land m + n + 5 \neq 0 \land m + n + 6 \neq 0$, then

$$\int \cos[e+fx]^6\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx \longrightarrow$$

$$\int (d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,(1-3\sin[e+fx]^2)\,dx + \frac{1}{d^4}\int (d\sin[e+fx])^{n+4}\,(a+b\sin[e+fx])^m\,(3-\sin[e+fx]^2)\,dx \longrightarrow$$

$$\frac{\cos[e+fx]\,(d\sin[e+fx])^{n+1}\,(a+b\sin[e+fx])^{m+1}}{a\,d\,f\,(n+1)} - \frac{b\,(m+n+2)\cos[e+fx]\,(d\sin[e+fx])^{n+2}\,(a+b\sin[e+fx])^{m+1}}{a^2\,d^2\,f\,(n+1)\,(n+2)} -$$
$$\frac{a\,(n+5)\cos[e+fx]\,(d\sin[e+fx])^{n+3}\,(a+b\sin[e+fx])^{m+1}}{b^2\,d^3\,f\,(m+n+5)\,(m+n+6)} + \frac{\cos[e+fx]\,(d\sin[e+fx])^{n+4}\,(a+b\sin[e+fx])^{m+1}}{b\,d^4\,f\,(m+n+6)} +$$
$$\frac{1}{a^2\,b^2\,d^2\,(n+1)\,(n+2)\,(m+n+5)\,(m+n+6)}\int (d\sin[e+fx])^{n+2}\,(a+b\sin[e+fx])^m\cdot$$
$$\big(a^4\,(n+1)\,(n+2)\,(n+3)\,(n+5) - a^2\,b^2\,(n+2)\,(2n+1)\,(m+n+5)\,(m+n+6) + b^4\,(m+n+2)\,(m+n+3)\,(m+n+5)\,(m+n+6) +$$
$$a\,b\,m\,\big(a^2\,(n+1)\,(n+2) - b^2\,(m+n+5)\,(m+n+6)\big)\sin[e+fx] -$$
$$\big(a^4\,(n+1)\,(n+2)\,(4+n)\,(n+5) + b^4\,(m+n+2)\,(m+n+4)\,(m+n+5)\,(m+n+6) - a^2\,b^2\,(n+1)\,(n+2)\,(m+n+5)\,(2n+2m+13)\big)\sin[e+fx]^2\big)\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^6*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Cos[e+f*x]*(d*Sin[e+f*x])^(n+1)*(a+b*Sin[e+f*x])^(m+1)/(a*d*f*(n+1)) -
  b*(m+n+2)*Cos[e+f*x]*(d*Sin[e+f*x])^(n+2)*(a+b*Sin[e+f*x])^(m+1)/(a^2*d^2*f*(n+1)*(n+2)) -
  a*(n+5)*Cos[e+f*x]*(d*Sin[e+f*x])^(n+3)*(a+b*Sin[e+f*x])^(m+1)/(b^2*d^3*f*(m+n+5)*(m+n+6)) +
  Cos[e+f*x]*(d*Sin[e+f*x])^(n+4)*(a+b*Sin[e+f*x])^(m+1)/(b*d^4*f*(m+n+6)) +
  1/(a^2*b^2*d^2*(n+1)*(n+2)*(m+n+5)*(m+n+6))*
    Int[(d*Sin[e+f*x])^(n+2)*(a+b*Sin[e+f*x])^m*
      Simp[a^4*(n+1)*(n+2)*(n+3)*(n+5)-a^2*b^2*(n+2)*(2*n+1)*(m+n+5)*(m+n+6)+b^4*(m+n+2)*(m+n+3)*(m+n+5)*(m+n+6) +
        a*b*m*(a^2*(n+1)*(n+2)-b^2*(m+n+5)*(m+n+6))*Sin[e+f*x] -
        (a^4*(n+1)*(n+2)*(4+n)*(n+5)+b^4*(m+n+2)*(m+n+4)*(m+n+5)*(m+n+6)-a^2*b^2*(n+1)*(n+2)*(m+n+5)*(2*n+2*m+13))*Sin[e+f*x]^2,x],x
FreeQ[{a,b,d,e,f,m,n},x] && NeQ[a^2-b^2,0] && IntegersQ[2*m,2*n] && NeQ[n,-1] && NeQ[n,-2] && NeQ[m+n+5,0] && NeQ[m+n+6,0] && Not[IG
```

3: $\int \cos[e+fx]^p\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx$ when $a^2-b^2\neq 0\wedge\left(m\mid 2n\mid \frac{p}{2}\right)\in\mathbb{Z}\wedge(m<-1\vee m==1\wedge p>0)$

Derivation: Algebraic expansion

Basis: $\cos[z]^2 == 1-\sin[z]^2$

Rule: If $a^2-b^2\neq 0\wedge\left(m\mid 2n\mid \frac{p}{2}\right)\in\mathbb{Z}\wedge(m<-1\vee m==1\wedge p>0)$, then

$$\int \cos[e+fx]^p\,(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,dx\to\int \text{ExpandTrig}\left[(d\sin[e+fx])^n\,(a+b\sin[e+fx])^m\,\left(1-\sin[e+fx]^2\right)^{p/2},x\right]dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Int[ExpandTrig[(d*sin[e+f*x])^n*(a+b*sin[e+f*x])^m*(1-sin[e+f*x]^2)^(p/2),x],x] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0] && IntegersQ[m,2*n,p/2] && (LtQ[m,-1] || EqQ[m,-1] && GtQ[p,0])
```

4. $\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2 \neq 0$

1: $\int \frac{(g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2 \neq 0 \land n \in \mathbb{Z} \land \left(n<0 \lor p+\frac{1}{2} \in \mathbb{Z}^+\right)$

Derivation: Algebraic expansion

Rule: If $a^2-b^2 \neq 0 \land n \in \mathbb{Z} \land \left(n<0 \lor p+\frac{1}{2} \in \mathbb{Z}^+\right)$, then

$$\int \frac{(g \operatorname{Cos}[e+f x])^p \operatorname{Sin}[e+f x]^n}{a+b \operatorname{Sin}[e+f x]} \, dx \to \int (g \operatorname{Cos}[e+f x])^p \operatorname{ExpandTrig}\left[\frac{\operatorname{Sin}[e+f x]^n}{a+b \operatorname{Sin}[e+f x]}, x\right] dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*sin[e_.+f_.*x_]^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  Int[ExpandTrig[(g*cos[e+f*x])^p,sin[e+f*x]^n/(a+b*sin[e+f*x]),x],x] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && IntegerQ[n] && (LtQ[n,0] || IGtQ[p+1/2,0])
```

2. $\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p>1$

1. $\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p>1 \land n<-1$

1: $\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p>1 \land n \leq -2$

Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Cos}[z]^2}{a+b \operatorname{Sin}[z]} == \frac{1}{a} - \frac{b \operatorname{Sin}[z]}{a^2} - \frac{(a^2-b^2) \operatorname{Sin}[z]^2}{a^2 (a+b \operatorname{Sin}[z])}$

Rule: If $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p>1 \land n \leq -2$, then

$$\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p\,(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx \longrightarrow$$

$$\frac{g^2}{a}\int (g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(d\,\mathrm{Sin}[e+f\,x])^n\,dx - \frac{b\,g^2}{a^2\,d}\int (g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(d\,\mathrm{Sin}[e+f\,x])^{n+1}\,dx - \frac{g^2\,(a^2-b^2)}{a^2\,d^2}\int \frac{(g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(d\,\mathrm{Sin}[e+f\,x])^{n+2}}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g^2/a*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^n,x] -
  b*g^2/(a^2*d)*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^(n+1),x] -
  g^2*(a^2-b^2)/(a^2*d^2)*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^(n+2)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && GtQ[p,1] && (LeQ[n,-2] || EqQ[n,-3/2] && EqQ[p,3/2])
```

**2:** $\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p\,(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\land\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\land\,p>1\,\land\,n<-1$

## Derivation: Algebraic expansion

Basis: $\frac{(g\,\mathrm{Cos}[z])^p\,(d\,\mathrm{Sin}[z])^n}{a+b\,\mathrm{Sin}[z]} == \frac{g^2\,(g\,\mathrm{Cos}[z])^{p-2}\,(d\,\mathrm{Sin}[z])^n\,(b-a\,\mathrm{Sin}[z])}{a\,b} + \frac{g^2\,(a^2-b^2)\,(g\,\mathrm{Cos}[z])^{p-2}\,(d\,\mathrm{Sin}[z])^{n+1}}{a\,b\,d\,(a+b\,\mathrm{Sin}[z])}$

Rule: If $a^2-b^2\neq 0\,\land\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\land\,p>1\,\land\,n<-1$, then

$$\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p\,(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx \longrightarrow$$

$$\frac{g^2}{a\,b}\int (g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(d\,\mathrm{Sin}[e+f\,x])^n\,(b-a\,\mathrm{Sin}[e+f\,x])\,dx + \frac{g^2\,(a^2-b^2)}{a\,b\,d}\int \frac{(g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(d\,\mathrm{Sin}[e+f\,x])^{n+1}}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g^2/(a*b)*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^n*(b-a*Sin[e+f*x]),x] +
  g^2*(a^2-b^2)/(a*b*d)*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^(n+1)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && GtQ[p,1] && (LtQ[n,-1] || EqQ[p,3/2] && EqQ[n,-1/2])
```

**2:** $\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p\,(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\wedge\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\wedge\,p>1$

## Derivation: Algebraic expansion

Basis: $\frac{(g\,\mathrm{Cos}[z])^p}{a+b\,\mathrm{Sin}[z]}==\frac{g^2\,(g\,\mathrm{Cos}[z])^{p-2}\,(a-b\,\mathrm{Sin}[z])}{b^2}-\frac{g^2\,(a^2-b^2)\,(g\,\mathrm{Cos}[z])^{p-2}}{b^2\,(a+b\,\mathrm{Sin}[z])}$

Rule: If $a^2-b^2\neq 0\,\wedge\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\wedge\,p>1$, then

$$\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p\,(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx\,\longrightarrow$$

$$\frac{g^2}{b^2}\int (g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(d\,\mathrm{Sin}[e+f\,x])^n\,(a-b\,\mathrm{Sin}[e+f\,x])\,dx-\frac{g^2\,(a^2-b^2)}{b^2}\int \frac{(g\,\mathrm{Cos}[e+f\,x])^{p-2}\,(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g^2/b^2*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^n*(a-b*Sin[e+f*x]),x] -
  g^2*(a^2-b^2)/b^2*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^n/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && GtQ[p,1]
```

**x:** $\int \frac{(g\,\mathrm{Cos}[e+f\,x])^p\,(d\,\mathrm{Sin}[e+f\,x])^n}{a+b\,\mathrm{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\wedge\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\wedge\,p>1$

## Derivation: Algebraic expansion

Basis: $(g\,\mathrm{Cos}[z])^p\,(d\,\mathrm{Sin}[z])^n==g^2\,(g\,\mathrm{Cos}[z])^{p-2}\,(d\,\mathrm{Sin}[z])^n-\frac{g^2\,(g\,\mathrm{Cos}[z])^{p-2}\,(d\,\mathrm{Sin}[z])^{n+2}}{d^2}$

Rule: If $a^2-b^2\neq 0\,\wedge\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\wedge\,p>1$, then

$$\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} dx \longrightarrow g^2 \int \frac{(g \operatorname{Cos}[e+f x])^{p-2} (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} dx - \frac{g^2}{d^2} \int \frac{(g \operatorname{Cos}[e+f x])^{p-2} (d \operatorname{Sin}[e+f x])^{n+2}}{a+b \operatorname{Sin}[e+f x]} dx$$

## Program code:

```
(* Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  g^2*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^n/(a+b*Sin[e+f*x]),x] -
  g^2/d^2*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^(n+2)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && GtQ[p,1] *)
```

3. $\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\wedge\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\wedge\,p<-1$

1: $\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\,\wedge\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\wedge\,p<-1\,\wedge\,n>1$

## Derivation: Algebraic expansion

Basis: $\frac{\text{Sin}[z]^2}{a+b\,\text{Sin}[z]}==\frac{a}{a^2-b^2}-\frac{b\,\text{Sin}[z]}{a^2-b^2}-\frac{a^2\,\text{Cos}[z]^2}{(a^2-b^2)\,(a+b\,\text{Sin}[z])}$

Rule: If $a^2-b^2\neq 0\,\wedge\,(2\,n\mid 2\,p)\in\mathbb{Z}\,\wedge\,p<-1\,\wedge\,n>1$, then

$$\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n}{a+b\,\text{Sin}[e+f\,x]}\,dx\,\longrightarrow$$

$$\frac{a\,d^2}{a^2-b^2}\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^{n-2}\,dx-\frac{b\,d}{a^2-b^2}\int (g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^{n-1}\,dx-\frac{a^2\,d^2}{g^2\,(a^2-b^2)}\int \frac{(g\,\text{Cos}[e+f\,x])^{p+2}\,(d\,\text{Sin}[e+f\,x])^{n-2}}{a+b\,\text{Sin}[e+f\,x]}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  a*d^2/(a^2-b^2)*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n-2),x] -
  b*d/(a^2-b^2)*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n-1),x] -
  a^2*d^2/(g^2*(a^2-b^2))*Int[(g*Cos[e+f*x])^(p+2)*(d*Sin[e+f*x])^(n-2)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && LtQ[p,-1] && GtQ[n,1]
```

**2:** $\int \frac{(g \cos[e+f x])^p (d \sin[e+f x])^n}{a+b \sin[e+f x]} dx$ when $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p<-1 \land n>0$

## Derivation: Algebraic expansion

Basis: $\frac{(g \cos[z])^p (d \sin[z])^n}{a+b \sin[z]} == -\frac{d (g \cos[z])^p (d \sin[z])^{n-1} (b-a \sin[z])}{a^2-b^2}+\frac{a b d (g \cos[z])^{p+2} (d \sin[z])^{n-1}}{g^2 (a^2-b^2) (a+b \sin[z])}$

Rule: If $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p<-1 \land n>0$, then

$$\int \frac{(g \cos[e+f x])^p (d \sin[e+f x])^n}{a+b \sin[e+f x]} dx \rightarrow$$

$$-\frac{d}{a^2-b^2} \int (g \cos[e+f x])^p (d \sin[e+f x])^{n-1} (b-a \sin[e+f x]) dx+\frac{a b d}{g^2 (a^2-b^2)} \int \frac{(g \cos[e+f x])^{p+2} (d \sin[e+f x])^{n-1}}{a+b \sin[e+f x]} dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -d/(a^2-b^2)*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n-1)*(b-a*Sin[e+f*x]),x] +
  a*b*d/(g^2*(a^2-b^2))*Int[(g*Cos[e+f*x])^(p+2)*(d*Sin[e+f*x])^(n-1)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && LtQ[p,-1] && GtQ[n,0]
```

**3:** $\int \frac{(g \cos[e+f x])^p (d \sin[e+f x])^n}{a+b \sin[e+f x]} dx$ when $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p<-1$

## Derivation: Algebraic expansion

Basis: $\frac{(g \cos[z])^p}{a+b \sin[z]} == \frac{g^2 (g \cos[z])^p (a-b \sin[z])}{g^2 (a^2-b^2)}-\frac{b^2 (g \cos[z])^{p+2}}{g^2 (a^2-b^2) (a+b \sin[z])}$

Rule: If $a^2-b^2 \neq 0 \land (2n \mid 2p) \in \mathbb{Z} \land p<-1$, then

$$\int \frac{(g \cos[e+f x])^p (d \sin[e+f x])^n}{a+b \sin[e+f x]} dx \rightarrow$$

$$\frac{1}{a^2-b^2}\int(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n\,(a-b\,\text{Sin}[e+f\,x])\,dx-\frac{b^2}{g^2\,(a^2-b^2)}\int\frac{(g\,\text{Cos}[e+f\,x])^{p+2}\,(d\,\text{Sin}[e+f\,x])^n}{a+b\,\text{Sin}[e+f\,x]}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  1/(a^2-b^2)*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^n*(a-b*Sin[e+f*x]),x] -
  b^2/(g^2*(a^2-b^2))*Int[(g*Cos[e+f*x])^(p+2)*(d*Sin[e+f*x])^n/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && LtQ[p,-1]
```

4. $\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2 \neq 0 \wedge (2n \mid 2p) \in \mathbb{Z} \wedge -1<p<1$

1. $\int \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]} \, (a+b \operatorname{Sin}[e+f x])} \, dx$ when $a^2-b^2 \neq 0$

1: $\int \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{\operatorname{Sin}[e+f x]} \, (a+b \operatorname{Sin}[e+f x])} \, dx$ when $a^2-b^2 \neq 0$

## Derivation: Integration by substitution

Basis: $\frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{\operatorname{Sin}[e+f x]} \, (a+b \operatorname{Sin}[e+f x])} == -\frac{4\sqrt{2}\, g}{f} \operatorname{Subst}\left[\frac{x^2}{\left((a+b)\, g^2+(a-b)\, x^4\right) \sqrt{1-\frac{x^4}{g^2}}}, x, \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{1+\operatorname{Sin}[e+f x]}}\right] \partial_x \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{1+\operatorname{Sin}[e+f x]}}$

## Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{\operatorname{Sin}[e+f x]} \, (a+b \operatorname{Sin}[e+f x])} \, dx \to -\frac{4\sqrt{2}\, g}{f} \operatorname{Subst}\left[\int \frac{x^2}{\left((a+b)\, g^2+(a-b)\, x^4\right) \sqrt{1-\frac{x^4}{g^2}}} \, dx, x, \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{1+\operatorname{Sin}[e+f x]}}\right]$$

## Program code:

```
Int[Sqrt[g_.*cos[e_.+f_.*x_]]/(Sqrt[sin[e_.+f_.*x_]]*(a_+b_.*sin[e_.+f_.*x_])),x_Symbol] :=
  -4*Sqrt[2]*g/f*Subst[Int[x^2/(((a+b)*g^2+(a-b)*x^4)*Sqrt[1-x^4/g^2]),x],x,Sqrt[g*Cos[e+f*x]]/Sqrt[1+Sin[e+f*x]]] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

**2:** $\int \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])} \, dx$ when $a^2-b^2 \neq 0$

## Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{\operatorname{Sin}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}} == 0$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])} \, dx \rightarrow \frac{\sqrt{\operatorname{Sin}[e+f x]}}{\sqrt{d \operatorname{Sin}[e+f x]}} \int \frac{\sqrt{g \operatorname{Cos}[e+f x]}}{\sqrt{\operatorname{Sin}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])} \, dx$$

## Program code:

```
Int[Sqrt[g_.*cos[e_.+f_.*x_]]/(Sqrt[d_*sin[e_.+f_.*x_]]*(a_+b_.*sin[e_.+f_.*x_])),x_Symbol] :=
  Sqrt[Sin[e+f*x]]/Sqrt[d*Sin[e+f*x]]*Int[Sqrt[g*Cos[e+f*x]]/(Sqrt[Sin[e+f*x]]*(a+b*Sin[e+f*x])),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0]
```

2. $\int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{g \operatorname{Cos}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])}\,dx$ when $a^2-b^2 \neq 0$

1: $\int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{\operatorname{Cos}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])}\,dx$ when $a^2-b^2 \neq 0$

## Derivation: Integration by substitution and algebraic expansion

Basis: $\frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{\operatorname{Cos}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])} == \frac{4\sqrt{2}\,d}{f} \operatorname{Subst}\left[\frac{x^2}{\left(a d^2+2 b d x^2+a x^4\right)\sqrt{1-\frac{x^4}{d^2}}},\,x,\,\frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{1+\operatorname{Cos}[e+f x]}}\right] \partial_x \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{1+\operatorname{Cos}[e+f x]}}$

Basis: Let $q \to \sqrt{-a^2+b^2}$, then $\frac{x^2}{a d^2+2 b d x^2+a x^4} == \frac{b+q}{2 q\,\left(d\,(b+q)+a x^2\right)} - \frac{b-q}{2 q\,\left(d\,(b-q)+a x^2\right)}$

Rule: If $a^2-b^2 \neq 0$, let $q \to \sqrt{-a^2+b^2}$, then

$$\int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{\operatorname{Cos}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])}\,dx \to \frac{4\sqrt{2}\,d}{f} \operatorname{Subst}\left[\int \frac{x^2}{\left(a d^2+2 b d x^2+a x^4\right)\sqrt{1-\frac{x^4}{d^2}}}\,dx,\,x,\,\frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{1+\operatorname{Cos}[e+f x]}}\right]$$

$$\to \frac{2\sqrt{2}\,d\,(b+q)}{f q} \operatorname{Subst}\left[\int \frac{1}{\left(d\,(b+q)+a x^2\right)\sqrt{1-\frac{x^4}{d^2}}}\,dx,\,x,\,\frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{1+\operatorname{Cos}[e+f x]}}\right] -$$

$$\frac{2\sqrt{2}\,d\,(b-q)}{f q} \operatorname{Subst}\left[\int \frac{1}{\left(d\,(b-q)+a x^2\right)\sqrt{1-\frac{x^4}{d^2}}}\,dx,\,x,\,\frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{1+\operatorname{Cos}[e+f x]}}\right]$$

## Program code:

```
Int[Sqrt[d_.*sin[e_.+f_.*x_]]/(Sqrt[cos[e_.+f_.*x_]]*(a_+b_.*sin[e_.+f_.*x_])),x_Symbol] :=
  With[{q=Rt[-a^2+b^2,2]},
  2*Sqrt[2]*d*(b+q)/(f*q)*Subst[Int[1/((d*(b+q)+a*x^2)*Sqrt[1-x^4/d^2]),x],x,Sqrt[d*Sin[e+f*x]]/Sqrt[1+Cos[e+f*x]]] -
   2*Sqrt[2]*d*(b-q)/(f*q)*Subst[Int[1/((d*(b-q)+a*x^2)*Sqrt[1-x^4/d^2]),x],x,Sqrt[d*Sin[e+f*x]]/Sqrt[1+Cos[e+f*x]]]] /;
FreeQ[{a,b,d,e,f},x] && NeQ[a^2-b^2,0]
```

**2:** $\int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{g \operatorname{Cos}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])}\,dx$ when $a^2-b^2 \neq 0$

## Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{\operatorname{Cos}[e+f x]}}{\sqrt{g \operatorname{Cos}[e+f x]}} == 0$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{g \operatorname{Cos}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])}\,dx \to \frac{\sqrt{\operatorname{Cos}[e+f x]}}{\sqrt{g \operatorname{Cos}[e+f x]}} \int \frac{\sqrt{d \operatorname{Sin}[e+f x]}}{\sqrt{\operatorname{Cos}[e+f x]}\,(a+b \operatorname{Sin}[e+f x])}\,dx$$

## Program code:

```
Int[Sqrt[d_.*sin[e_.+f_.*x_]]/(Sqrt[g_.*cos[e_.+f_.*x_]]*(a_+b_.*sin[e_.+f_.*x_])),x_Symbol] :=
  Sqrt[Cos[e+f*x]]/Sqrt[g*Cos[e+f*x]]*Int[Sqrt[d*Sin[e+f*x]]/(Sqrt[Cos[e+f*x]]*(a+b*Sin[e+f*x])),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0]
```

**3:** $\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n}{a+b\,\text{Sin}[e+f\,x]}\,dx$ when $a^2-b^2\neq 0\wedge(2\,n\mid 2\,p)\in\mathbb{Z}\wedge -1<p<1\wedge n>0$

## Derivation: Algebraic expansion

Basis: $\frac{(d\,z)^n}{a+b\,z}=\frac{d\,(d\,z)^{n-1}}{b}-\frac{a\,d\,(d\,z)^{n-1}}{b\,(a+b\,z)}$

Rule: If $a^2-b^2\neq 0\wedge(2\,n\mid 2\,p)\in\mathbb{Z}\wedge -1<p<1\wedge n>0$, then

$$\int \frac{(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^n}{a+b\,\text{Sin}[e+f\,x]}\,dx\to\frac{d}{b}\int(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^{n-1}\,dx-\frac{a\,d}{b}\int\frac{(g\,\text{Cos}[e+f\,x])^p\,(d\,\text{Sin}[e+f\,x])^{n-1}}{a+b\,\text{Sin}[e+f\,x]}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  d/b*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n-1),x] -
  a*d/b*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n-1)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && LtQ[-1,p,1] && GtQ[n,0]
```

4: $\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx$ when $a^2-b^2 \neq 0 \wedge (2n \mid 2p) \in \mathbb{Z} \wedge -1<p<1 \wedge n<0$

## Derivation: Algebraic expansion

Basis: $\frac{(d z)^n}{a+b z} = \frac{(d z)^n}{a} - \frac{b (d z)^{n+1}}{a d (a+b z)}$

Rule: If $a^2-b^2 \neq 0 \wedge (2n \mid 2p) \in \mathbb{Z} \wedge -1<p<1 \wedge n<0$, then

$$\int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n}{a+b \operatorname{Sin}[e+f x]} \, dx \longrightarrow \frac{1}{a} \int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n \, dx - \frac{b}{a d} \int \frac{(g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^{n+1}}{a+b \operatorname{Sin}[e+f x]} \, dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  1/a*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^n,x] -
  b/(a*d)*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n+1)/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[2*n,2*p] && LtQ[-1,p,1] && LtQ[n,0]
```

5. $\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land m \in \mathbb{Z} \land (m>0 \lor n \in \mathbb{Z})$

1: $\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^2 \, dx$ when $a^2-b^2 \neq 0$

## Derivation: Algebraic expansion

## Rule: If $a^2-b^2 \neq 0$, then

$$\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^2 \, dx \longrightarrow$$

$$\frac{2 a b}{d} \int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^{n+1} \, dx + \int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n \left(a^2+b^2 \operatorname{Sin}[e+f x]^2\right) dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^2,x_Symbol] :=
  2*a*b/d*Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^(n+1),x] +
  Int[(g*Cos[e+f*x])^p*(d*Sin[e+f*x])^n*(a^2+b^2*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,d,e,f,g,n,p},x] && NeQ[a^2-b^2,0]
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \wedge m \in \mathbb{Z} \wedge (m>0 \vee n \in \mathbb{Z})$

## Derivation: Algebraic expansion

## Rule: If $a^2-b^2 \neq 0 \wedge m \in \mathbb{Z} \wedge (m>0 \vee n \in \mathbb{Z})$, then

$$\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx \rightarrow \int (g \operatorname{Cos}[e+f x])^p \operatorname{ExpandTrig}\left[(d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m, x\right] dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  Int[ExpandTrig[(g*cos[e+f*x])^p,(d*sin[e+f*x])^n*(a+b*sin[e+f*x])^m,x],x] /;
FreeQ[{a,b,d,e,f,g,n,p},x] && NeQ[a^2-b^2,0] && IntegerQ[m] && (GtQ[m,0] || IntegerQ[n])
```

**6:** $\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx$ when $a^2-b^2 \neq 0 \land (m \mid 2n \mid 2p) \in \mathbb{Z} \land m<0 \land p>1 \land n \leq -2$

## Derivation: Algebraic expansion

Basis: $\operatorname{Cos}[z]^2 = \frac{a+b \operatorname{Sin}[z]}{a} - \frac{b \operatorname{Sin}[z] (a+b \operatorname{Sin}[z])}{a^2} - \frac{(a^2-b^2) \operatorname{Sin}[z]^2}{a^2}$

Rule: If $a^2-b^2 \neq 0 \land (m \mid 2n \mid 2p) \in \mathbb{Z} \land m<0 \land p>1 \land n \leq -2$, then

$$\int (g \operatorname{Cos}[e+f x])^p (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^m \, dx \to$$

$$\frac{g^2}{a} \int (g \operatorname{Cos}[e+f x])^{p-2} (d \operatorname{Sin}[e+f x])^n (a+b \operatorname{Sin}[e+f x])^{m+1} \, dx -$$
$$\frac{b g^2}{a^2 d} \int (g \operatorname{Cos}[e+f x])^{p-2} (d \operatorname{Sin}[e+f x])^{n+1} (a+b \operatorname{Sin}[e+f x])^{m+1} \, dx -$$
$$\frac{g^2 (a^2-b^2)}{a^2 d^2} \int (g \operatorname{Cos}[e+f x])^{p-2} (d \operatorname{Sin}[e+f x])^{n+2} (a+b \operatorname{Sin}[e+f x])^m \, dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(d_.*sin[e_.+f_.*x_])^n_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g^2/a*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^n*(a+b*Sin[e+f*x])^(m+1),x] -
  b*g^2/(a^2*d)*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^(n+1)*(a+b*Sin[e+f*x])^(m+1),x] -
  g^2*(a^2-b^2)/(a^2*d^2)*Int[(g*Cos[e+f*x])^(p-2)*(d*Sin[e+f*x])^(n+2)*(a+b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,d,e,f,g},x] && NeQ[a^2-b^2,0] && IntegersQ[m,2*n,2*p] && LtQ[m,0] && GtQ[p,1] && (LeQ[n,-2] || EqQ[m,-1] && EqQ[n,-3/2] &
```

8. $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $a^2-b^2=0$

**1:** $\int \operatorname{Cos}[e+f x]^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $a^2-b^2=0 \land m \in \mathbb{Z} \land 2m+p=0$

Derivation: Algebraic simplification

Basis: If $a^2-b^2=0 \land m \in \mathbb{Z} \land 2m+p=0$, then $\operatorname{Cos}[z]^p (a+b \operatorname{Sin}[z])^m = \frac{a^{2m}}{(a-b \operatorname{Sin}[z])^m}$

Rule: If $a^2-b^2=0 \land m \in \mathbb{Z} \land 2m+p=0$, then

$$\int \operatorname{Cos}[e+f x]^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx \to a^{2m} \int \frac{(c+d \operatorname{Sin}[e+f x])^n}{(a-b \operatorname{Sin}[e+f x])^m} \, dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^(2*m)*Int[(c+d*Sin[e+f*x])^n/(a-b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[a^2-b^2,0] && IntegersQ[m,p] && EqQ[2*m+p,0]
```

**2:** $\int (g \operatorname{Cos}[e+f x])^p (a+b \operatorname{Sin}[e+f x])^m (c+d \operatorname{Sin}[e+f x])^n \, dx$ when $a^2-b^2=0 \land m \in \mathbb{Z} \land (2m+p=0 \lor 2m+p>0 \land p<-1)$

Derivation: Algebraic expansion

Basis: If $a^2-b^2=0$, then $a+b \operatorname{Sin}[z] = \frac{a^2 (g \operatorname{Cos}[z])^2}{g^2 (a-b \operatorname{Sin}[z])}$

Note: By making the degree of the cosine factor in the integrand nonnegative, this rule removes the removable singularities from the integrand and hence from the resulting antiderivatives.

Rule: If $a^2-b^2=0 \land m \in \mathbb{Z} \land (2m+p=0 \lor 2m+p>0 \land p<-1)$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx \;\rightarrow\; \frac{a^{2\,m}}{g^{2\,m}}\int\frac{(g\,\mathrm{Cos}[e+f\,x])^{2\,m+p}\,(c+d\,\mathrm{Sin}[e+f\,x])^n}{(a-b\,\mathrm{Sin}[e+f\,x])^m}\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  (a/g)^(2*m)*Int[(g*Cos[e+f*x])^(2*m+p)*(c+d*Sin[e+f*x])^n/(a-b*Sin[e+f*x])^m,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[a^2-b^2,0] && IntegerQ[m] && (EqQ[2*m+p,0] || GtQ[2*m+p,0] && LtQ[p,-1])
```

3. $\int \mathrm{Cos}[e+f\,x]^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0 \land \frac{p}{2}\in\mathbb{Z}$

1: $\int \mathrm{Cos}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0 \land (2\,m\mid 2\,n)\in\mathbb{Z}$

## Derivation: Algebraic simplification

Basis: If $a^2-b^2==0$, then $\mathrm{Cos}[z]^2==\frac{1}{b^2}\,(a+b\,\mathrm{Sin}[z])\,(a-b\,\mathrm{Sin}[z])$

Rule: If $a^2-b^2==0 \land (2\,m\mid 2\,n)\in\mathbb{Z}$, then

$$\int \mathrm{Cos}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx \;\rightarrow\; \frac{1}{b^2}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(a-b\,\mathrm{Sin}[e+f\,x])\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  1/b^2*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*(a-b*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && IntegersQ[2*m,2*n]
```

**2:** $\int \cos[e+fx]^p\,(a+b\sin[e+fx])^m\,(c+d\sin[e+fx])^n\,dx$ when $a^2-b^2=0\,\land\,\frac{p}{2}\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$

Derivation: Algebraic expansion, piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0\,\land\,\frac{p}{2}\in\mathbb{Z}$, then $\cos[z]^p = a^{-p}\,(a+b\sin[z])^{p/2}\,(a-b\sin[z])^{p/2}$

Basis: $\partial_x \frac{\cos[e+fx]}{\sqrt{1+\sin[e+fx]}\,\sqrt{1-\sin[e+fx]}} = 0$

Basis: $\cos[e+fx] = \frac{1}{f}\,\partial_x \sin[e+fx]$

Rule: If $a^2-b^2=0\,\land\,\frac{p}{2}\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$, then

$$\int \cos[e+fx]^p\,(a+b\sin[e+fx])^m\,(c+d\sin[e+fx])^n\,dx\,\longrightarrow$$

$$a^{-p}\int (a+b\sin[e+fx])^{m+p/2}\,(a-b\sin[e+fx])^{p/2}\,(c+d\sin[e+fx])^n\,dx\,\longrightarrow$$

$$\frac{a^m\cos[e+fx]}{\sqrt{1+\sin[e+fx]}\,\sqrt{1-\sin[e+fx]}}\int \cos[e+fx]\left(1+\frac{b}{a}\sin[e+fx]\right)^{m+\frac{p-1}{2}}\left(1-\frac{b}{a}\sin[e+fx]\right)^{\frac{p-1}{2}}(c+d\sin[e+fx])^n\,dx\,\longrightarrow$$

$$\frac{a^m\cos[e+fx]}{f\sqrt{1+\sin[e+fx]}\,\sqrt{1-\sin[e+fx]}}\,\mathrm{Subst}\left[\int\left(1+\frac{b}{a}x\right)^{m+\frac{p-1}{2}}\left(1-\frac{b}{a}x\right)^{\frac{p-1}{2}}(c+dx)^n\,dx,\,x,\,\sin[e+fx]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^m*Cos[e+f*x]/(f*Sqrt[1+Sin[e+f*x]]*Sqrt[1-Sin[e+f*x]])*
    Subst[Int[(1+b/a*x)^(m+(p-1)/2)*(1-b/a*x)^((p-1)/2)*(c+d*x)^n,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[a^2-b^2,0] && IntegerQ[p/2] && IntegerQ[m]
```

**3:** $\int \text{Cos}[e+f x]^p (a+b \text{Sin}[e+f x])^m (c+d \text{Sin}[e+f x])^n \, dx$ when $a^2-b^2=0 \land \frac{p}{2} \in \mathbb{Z} \land m \notin \mathbb{Z}$

Derivation: Algebraic expansion, piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0 \land \frac{p}{2} \in \mathbb{Z}$, then $\text{Cos}[z]^p = a^{-p} (a+b \text{Sin}[z])^{p/2} (a-b \text{Sin}[z])^{p/2}$

Basis: If $a^2-b^2=0$, then $\partial_x \frac{\text{Cos}[e+f x]}{\sqrt{a+b \text{Sin}[e+f x]} \sqrt{a-b \text{Sin}[e+f x]}} = 0$

Basis: $\text{Cos}[e+f x] = \frac{1}{f} \partial_x \text{Sin}[e+f x]$

Rule: If $a^2-b^2=0 \land \frac{p}{2} \in \mathbb{Z} \land m \notin \mathbb{Z}$, then

$$\int \text{Cos}[e+f x]^p (a+b \text{Sin}[e+f x])^m (c+d \text{Sin}[e+f x])^n \, dx \rightarrow$$

$$a^{-p} \int (a+b \text{Sin}[e+f x])^{m+p/2} (a-b \text{Sin}[e+f x])^{p/2} (c+d \text{Sin}[e+f x])^n \, dx \rightarrow$$

$$\frac{\text{Cos}[e+f x]}{a^{p-2} \sqrt{a+b \text{Sin}[e+f x]} \sqrt{a-b \text{Sin}[e+f x]}} \int \text{Cos}[e+f x] (a+b \text{Sin}[e+f x])^{m+\frac{p}{2}-\frac{1}{2}} (a-b \text{Sin}[e+f x])^{\frac{p}{2}-\frac{1}{2}} (c+d \text{Sin}[e+f x])^n \, dx \rightarrow$$

$$\frac{\text{Cos}[e+f x]}{a^{p-2} f \sqrt{a+b \text{Sin}[e+f x]} \sqrt{a-b \text{Sin}[e+f x]}} \text{Subst}\left[\int (a+b x)^{m+\frac{p}{2}-\frac{1}{2}} (a-b x)^{\frac{p}{2}-\frac{1}{2}} (c+d x)^n \, dx, x, \text{Sin}[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Cos[e+f*x]/(a^(p-2)*f*Sqrt[a+b*Sin[e+f*x]]*Sqrt[a-b*Sin[e+f*x]])*
    Subst[Int[(a+b*x)^(m+p/2-1/2)*(a-b*x)^(p/2-1/2)*(c+d*x)^n,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[a^2-b^2,0] && IntegerQ[p/2] && Not[IntegerQ[m]]
```

**4:** $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $a^2-b^2==0\,\land\,m\in\mathbb{Z}^+$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\to\,\int (g\,\text{Cos}[e+f\,x])^p\,\text{ExpandTrig}\big[(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n,\,x\big]\,dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(g*cos[e+f*x])^p,(a+b*sin[e+f*x])^m*(c+d*sin[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[a^2-b^2,0] && IGtQ[m,0] && (IntegerQ[p] || IGtQ[n,0])
```

5. $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0$

**1:** $\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\,\frac{(g\,\text{Cos}[e+f\,x])^{p-1}}{(1+\text{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\text{Sin}[e+f\,x])^{\frac{p-1}{2}}}==0$

Basis: $\text{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\,\text{Sin}[e+f\,x]$

Rule: If $a^2-b^2==0\,\land\,m\in\mathbb{Z}$, then

$$\int (g\,\text{Cos}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\to$$

$$\frac{a^m g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}}{(1+\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}}\int \mathrm{Cos}[e+f\,x]\,\left(1+\frac{b}{a}\,\mathrm{Sin}[e+f\,x]\right)^{m+\frac{p-1}{2}}\left(1-\frac{b}{a}\,\mathrm{Sin}[e+f\,x]\right)^{\frac{p-1}{2}}(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$\frac{a^m g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}}{f\,(1+\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(1-\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}}\,\mathrm{Subst}\left[\int\left(1+\frac{b}{a}\,x\right)^{m+\frac{p-1}{2}}\left(1-\frac{b}{a}\,x\right)^{\frac{p-1}{2}}(c+d\,x)^n\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^m*g*(g*Cos[e+f*x])^(p-1)/(f*(1+Sin[e+f*x])^((p-1)/2)*(1-Sin[e+f*x])^((p-1)/2))*
    Subst[Int[(1+b/a*x)^(m+(p-1)/2)*(1-b/a*x)^((p-1)/2)*(c+d*x)^n,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[a^2-b^2,0] && IntegerQ[m]
```

**2:** $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2==0\,\land\,m\notin\mathbb{Z}$

## Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0$, then $\partial_x\frac{\mathrm{Cos}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sin}[e+f\,x]}}==0$

Basis: $\mathrm{Cos}[e+f\,x]==\frac{1}{f}\,\partial_x\,\mathrm{Sin}[e+f\,x]$

Rule: If $a^2-b^2==0\,\land\,m\notin\mathbb{Z}$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$\frac{g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}}{(a+b\,\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(a-b\,\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}}\int \mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+\frac{p-1}{2}}\,(a-b\,\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\;\longrightarrow$$

$$\frac{g\,(g\,\mathrm{Cos}[e+f\,x])^{p-1}}{f\,(a+b\,\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}\,(a-b\,\mathrm{Sin}[e+f\,x])^{\frac{p-1}{2}}}\,\mathrm{Subst}\Big[\int(a+b\,x)^{m+\frac{p-1}{2}}\,(a-b\,x)^{\frac{p-1}{2}}\,(c+d\,x)^n\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\Big]$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  g*(g*Cos[e+f*x])^(p-1)/(f*(a+b*Sin[e+f*x])^((p-1)/2)*(a-b*Sin[e+f*x])^((p-1)/2))*
    Subst[Int[(a+b*x)^(m+(p-1)/2)*(a-b*x)^((p-1)/2)*(c+d*x)^n,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]]
```

9. $\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0$

   1. $\int(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\,\land\,\frac{p}{2}\in\mathbb{Z}^+$

      1: $\int\mathrm{Cos}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0$

## Derivation: Algebraic expansion

## Basis: $\mathrm{Cos}[z]^2 == 1-\mathrm{Sin}[z]^2$

## Rule: If $a^2-b^2\neq 0$, then

$$\int\mathrm{Cos}[e+f\,x]^2\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx\,\rightarrow\,\int(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(1-\mathrm{Sin}[e+f\,x]^2\right)\,dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^2*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*(1-Sin[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[a^2-b^2,0] && (IGtQ[m,0] || IntegersQ[2*m,2*n])
```

**2:** $\int \text{Cos}[e+f x]^p \,(a+b\,\text{Sin}[e+f x])^m \,(c+d\,\text{Sin}[e+f x])^n \,dx$ when $a^2-b^2\neq 0 \wedge \frac{p}{2}\in\mathbb{Z}^+$

## Derivation: Algebraic expansion

## Basis: $\text{Cos}[z]^2 == 1-\text{Sin}[z]^2$

## Rule: If $a^2-b^2\neq 0 \wedge \frac{p}{2}\in\mathbb{Z}^+$, then

$$\int \text{Cos}[e+f x]^p \,(a+b\,\text{Sin}[e+f x])^m \,(c+d\,\text{Sin}[e+f x])^n \,dx \longrightarrow$$
$$\int \text{ExpandTrig}\left[(a+b\,\text{Sin}[e+f x])^m \,(c+d\,\text{Sin}[e+f x])^n \,\left(1-\text{Sin}[e+f x]^2\right)^{p/2},\,x\right] dx$$

## Program code:

```
Int[cos[e_.+f_.*x_]^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(a+b*sin[e+f*x])^m*(c+d*sin[e+f*x])^n*(1-sin[e+f*x]^2)^(p/2),x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[a^2-b^2,0] && IGtQ[p/2,0] && (IGtQ[m,0] || IntegersQ[2*m,2*n])
```

**2:** $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0\ \wedge\ (m\mid n)\in\mathbb{Z}$

## Derivation: Algebraic expansion

## Rule: If $a^2-b^2\neq 0\ \wedge\ (m\mid n)\in\mathbb{Z}$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx \to \int \mathrm{ExpandTrig}\left[(g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n,\ x\right]dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_*(c_+d_.*sin[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(g*cos[e+f*x])^p*(a+b*sin[e+f*x])^m*(c+d*sin[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[a^2-b^2,0] && IntegersQ[2*m,2*n]
```

**X:** $\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$ when $a^2-b^2\neq 0$

## Rule: If $a^2-b^2\neq 0$, then

$$\int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx \to \int (g\,\mathrm{Cos}[e+f\,x])^p\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,dx$$

## Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(g*Cos[e+f*x])^p*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && NeQ[a^2-b^2,0]
```

# Rules for integrands of the form $(g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n$

**1:** $\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((g\,\text{Cos}[e+f\,x])^p\,(g\,\text{Sec}[e+f\,x])^p\right)==0$

Rule: If $p\notin\mathbb{Z}$, then

$$\int (g\,\text{Sec}[e+f\,x])^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\rightarrow$$
$$g^{2\,\text{IntPart}[p]}\,(g\,\text{Cos}[e+f\,x])^{\text{FracPart}[p]}\,(g\,\text{Sec}[e+f\,x])^{\text{FracPart}[p]}\int\frac{(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n}{(g\,\text{Cos}[e+f\,x])^p}\,dx$$

Program code:

```
Int[(g_.*sec[e_.+f_.*x_])^p_*(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^(2*IntPart[p])*(g*Cos[e+f*x])^FracPart[p]*(g*Sec[e+f*x])^FracPart[p]*
    Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n/(g*Cos[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && Not[IntegerQ[p]]
```

```
Int[(g_.*csc[e_.+f_.*x_])^p_*(a_.+b_.*cos[e_.+f_.*x_])^m_.*(c_.+d_.*cos[e_.+f_.*x_])^n_.,x_Symbol] :=
  g^(2*IntPart[p])*(g*Sin[e+f*x])^FracPart[p]*(g*Csc[e+f*x])^FracPart[p]*
    Int[(a+b*Cos[e+f*x])^m*(c+d*Cos[e+f*x])^n/(g*Sin[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && Not[IntegerQ[p]]
```