Rules for integrands of the form $u\,\text{Hyper}[d\,(a+b\,\text{Log}[c\,x^n])]^p$

1. $\int u\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

   1. $\int \text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

      1: $\int \text{Sinh}[b\,\text{Log}[c\,x^n]]^p\,dx$

Derivation: Algebraic simplification

Basis: $\text{Sinh}[b\,\text{Log}[c\,x^n]] = \frac{1}{2}\,(c\,x^n)^b - \frac{1}{2\,(c\,x^n)^b}$

Basis: $\text{Cosh}[b\,\text{Log}[c\,x^n]] = \frac{1}{2}\,(c\,x^n)^b + \frac{1}{2\,(c\,x^n)^b}$

Rule:

$$\int \text{Sinh}[b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow \int \left(\frac{(c\,x^n)^b}{2} - \frac{1}{2\,(c\,x^n)^b}\right)^p dx$$

Program code:

```
Int[Sinh[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[((c*x^n)^b/2 - 1/(2*(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[b,n,p]
```

```
Int[Cosh[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[((c*x^n)^b/2 + 1/(2*(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[b,n,p]
```

1. $\int \text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$ when $p \in \mathbb{Z}^+ \land b^2\,d^2\,n^2\,p^2-1 \neq 0$

   1: $\int \text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$ when $b^2\,d^2\,n^2-1 \neq 0$

Rule: If $b^2\,d^2\,n^2-1 \neq 0$, then

$$\int \text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx \longrightarrow -\frac{x\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]}{b^2\,d^2\,n^2-1}+\frac{b\,d\,n\,x\,\text{Cosh}[d\,(a+b\,\text{Log}[c\,x^n])]}{b^2\,d^2\,n^2-1}$$

Program code:

```
Int[Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  -x*Sinh[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2-1) +
  b*d*n*x*Cosh[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2-1) /;
FreeQ[{a,b,c,d,n},x] && NeQ[b^2*d^2*n^2-1,0]
```

```
Int[Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  -x*Cosh[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2-1) +
  b*d*n*x*Sinh[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2-1) /;
FreeQ[{a,b,c,d,n},x] && NeQ[b^2*d^2*n^2-1,0]
```

2: $\int \text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p\,dx$ when $p-1\in\mathbb{Z}^+\wedge\,b^2\,d^2\,n^2\,p^2-1\neq 0$

Rule: If $p-1\in\mathbb{Z}^+\wedge\,b^2\,d^2\,n^2\,p^2-1\neq 0$, then

$$\int \text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p\,dx\,\longrightarrow$$

$$-\frac{x\,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p}{b^2\,d^2\,n^2\,p^2-1}+\frac{b\,d\,n\,p\,x\,\text{Cosh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^{p-1}}{b^2\,d^2\,n^2\,p^2-1}-\frac{b^2\,d^2\,n^2\,p\,(p-1)}{b^2\,d^2\,n^2\,p^2-1}\int \text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^{p-2}\,dx$$

Program code:

```
Int[Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  -x*Sinh[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*n^2*p^2-1) +
  b*d*n*p*x*Cosh[d*(a+b*Log[c*x^n])]*Sinh[d*(a+b*Log[c*x^n])]^(p-1)/(b^2*d^2*n^2*p^2-1) -
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2-1)*Int[Sinh[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2-1,0]
```

```
Int[Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  -x*Cosh[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*n^2*p^2-1) +
  b*d*n*p*x*Cosh[d*(a+b*Log[c*x^n])]^(p-1)*Sinh[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2*p^2-1) +
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2-1)*Int[Cosh[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2-1,0]
```

2. $\int \text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx$

1: $\int \text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}^+\land b^2\,d^2\,p^2-1==0$

Derivation: Algebraic expansion

Basis: If $b^2\,d^2\,p^2-1==0\land p\in\mathbb{Z}$, then $\text{Sinh}[d\,(a+b\,\text{Log}[x])]^p==\frac{1}{2^p\,b^p\,d^p\,p^p}\left(-e^{-a\,b\,d^2\,p}\,x^{-\frac{1}{p}}+e^{a\,b\,d^2\,p}\,x^{\frac{1}{p}}\right)^p$

Basis: If $b^2\,d^2\,p^2-1==0\land p\in\mathbb{Z}$, then $\text{Cosh}[d\,(a+b\,\text{Log}[x])]^p==\frac{1}{2^p}\left(e^{-a\,b\,d^2\,p}\,x^{-\frac{1}{p}}+e^{a\,b\,d^2\,p}\,x^{\frac{1}{p}}\right)^p$

Note: The above identities need to be formally derived, and possibly the domain of p expanded.

Rule: If $p\in\mathbb{Z}^+\land b^2\,d^2\,p^2-1==0$, then

$$\int \text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx\to\frac{1}{2^p\,b^p\,d^p\,p^p}\int \text{ExpandIntegrand}\left[\left(-e^{-a\,b\,d^2\,p}\,x^{-\frac{1}{p}}+e^{a\,b\,d^2\,p}\,x^{\frac{1}{p}}\right)^p,\,x\right]dx$$

Program code:

```
Int[Sinh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/(2^p*b^p*d^p*p^p)*Int[ExpandIntegrand[(-E^(-a*b*d^2*p)*x^(-1/p)+E^(a*b*d^2*p)*x^(1/p))^p,x],x] /;
FreeQ[{a,b,d},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2-1,0]
```

```
Int[Cosh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/2^p*Int[ExpandIntegrand[(E^(-a*b*d^2*p)*x^(-1/p)+E^(a*b*d^2*p)*x^(1/p))^p,x],x] /;
FreeQ[{a,b,d},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2-1,0]
```

x: $\int \text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[d\,(a+b\,\text{Log}[x])] == \frac{e^{a\,d}}{2}\,x^{b\,d}\,\left(1-e^{-2\,a\,d}\,x^{-2\,b\,d}\right)$

Basis: $\text{Cosh}[d\,(a+b\,\text{Log}[x])] == \frac{e^{a\,d}}{2}\,x^{b\,d}\,\left(1+e^{-2\,a\,d}\,x^{-2\,b\,d}\right)$

Rule: If $p\in\mathbb{Z}$, then

$$\int \text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx \to \frac{e^{a\,d\,p}}{2^p}\int x^{b\,d\,p}\,\left(1-e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p\,dx$$

Program code:

```
(* Int[Sinh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  E^(a*d*p)/2^p*Int[x^(b*d*p)*(1-1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d},x] && IntegerQ[p] *)
```

```
(* Int[Cosh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  E^(a*d*p)/2^p*Int[x^(b*d*p)*(1+1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d},x] && IntegerQ[p] *)
```

2: $\int \text{Sinh}[d\,(a + b\,\text{Log}[x])]^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\text{Sinh}[d\,(a+b\,\text{Log}[x])]^p}{x^{b\,d\,p}\,\left(1-e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p} == 0$

Basis: $\partial_x \frac{\text{Cosh}[d\,(a+b\,\text{Log}[x])]^p}{x^{b\,d\,p}\,\left(1+e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int \text{Sinh}[d\,(a + b\,\text{Log}[x])]^p\,dx \longrightarrow \frac{\text{Sinh}[d\,(a + b\,\text{Log}[x])]^p}{x^{b\,d\,p}\,\left(1 - e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p} \int x^{b\,d\,p}\,\left(1 - e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p\,dx$$

Program code:

```
Int[Sinh[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Sinh[d*(a+b*Log[x])]^p/(x^(b*d*p)*(1-1/(E^(2*a*d)*x^(2*b*d)))^p)*
    Int[x^(b*d*p)*(1-1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

```
Int[Cosh[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Cosh[d*(a+b*Log[x])]^p/(x^(b*d*p)*(1+1/(E^(2*a*d)*x^(2*b*d)))^p)*
    Int[x^(b*d*p)*(1+1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

3: $\int \text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\partial_x\,(c\,x^n)$

Rule:

$$\int \text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx \to \frac{x}{(c\,x^n)^{1/n}}\int \frac{(c\,x^n)^{1/n}\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p}{x}\,dx$$

$$\to \frac{x}{n\,(c\,x^n)^{1/n}}\,\text{Subst}\left[\int x^{1/n-1}\,\text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx,\,x,\,c\,x^n\right]$$

```
Int[Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Sinh[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Cosh[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

2. $\int (e\,x)^m\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

   1. $\int (e\,x)^m\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$ when $p \in \mathbb{Z}^+ \land b^2\,d^2\,n^2\,p^2-(m+1)^2 \neq 0$

      1: $\int (e\,x)^m\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$ when $b^2\,d^2\,n^2-(m+1)^2 \neq 0$

Rule: If $b^2\,d^2\,n^2-(m+1)^2 \neq 0$, then

$$\int (e\,x)^m \,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right] \,dx \,\rightarrow\, -\frac{(m+1)\,(e\,x)^{m+1}\,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]}{b^2\,d^2\,e\,n^2-e\,(m+1)^2}+\frac{b\,d\,n\,(e\,x)^{m+1}\,\text{Cosh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]}{b^2\,d^2\,e\,n^2-e\,(m+1)^2}$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  -(m+1)*(e*x)^(m+1)*Sinh[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2-e*(m+1)^2) +
  b*d*n*(e*x)^(m+1)*Cosh[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2-e*(m+1)^2) /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b^2*d^2*n^2-(m+1)^2,0]
```

```
Int[(e_.*x_)^m_.*Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  -(m+1)*(e*x)^(m+1)*Cosh[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2-e*(m+1)^2) +
  b*d*n*(e*x)^(m+1)*Sinh[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2-e*(m+1)^2) /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b^2*d^2*n^2-(m+1)^2,0]
```

2: $\int (e\,x)^m \,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p \,dx$ when $p-1\in\mathbb{Z}^+ \land\, b^2\,d^2\,n^2\,p^2-(m+1)^2\neq 0$

Rule: If $p-1\in\mathbb{Z}^+ \land\, b^2\,d^2\,n^2\,p^2-(m+1)^2\neq 0$, then

$$\int (e\,x)^m \,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p \,dx \,\rightarrow$$

$$-\frac{(m+1)\,(e\,x)^{m+1}\,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p}{b^2\,d^2\,e\,n^2\,p^2-e\,(m+1)^2}+\frac{b\,d\,n\,p\,(e\,x)^{m+1}\,\text{Cosh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^{p-1}}{b^2\,d^2\,e\,n^2\,p^2-e\,(m+1)^2}-$$
$$\frac{b^2\,d^2\,n^2\,p\,(p-1)}{b^2\,d^2\,n^2\,p^2-(m+1)^2}\int (e\,x)^m \,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^{p-2} \,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  -(m+1)*(e*x)^(m+1)*Sinh[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*e*n^2*p^2-e*(m+1)^2) +
  b*d*n*p*(e*x)^(m+1)*Cosh[d*(a+b*Log[c*x^n])]*Sinh[d*(a+b*Log[c*x^n])]^(p-1)/(b^2*d^2*e*n^2*p^2-e*(m+1)^2) -
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2-(m+1)^2)*Int[(e*x)^m*Sinh[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2-(m+1)^2,0]
```

```
Int[(e_.*x_)^m_.*Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  -(m+1)*(e*x)^(m+1)*Cosh[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*e*n^2*p^2-e*(m+1)^2) +
  b*d*n*p*(e*x)^(m+1)*Sinh[d*(a+b*Log[c*x^n])]*Cosh[d*(a+b*Log[c*x^n])]^(p-1)/(b^2*d^2*e*n^2*p^2-e*(m+1)^2) +
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2-(m+1)^2)*Int[(e*x)^m*Cosh[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2-(m+1)^2,0]
```

2. $\int (e\,x)^m \,\text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx$

1: $\int (e\,x)^m \,\text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}^+ \land b^2\,d^2\,p^2-(m+1)^2==0$

Derivation: Algebraic expansion

Basis: If $b^2\,d^2\,p^2-(m+1)^2==0 \land p\in\mathbb{Z}$, then $\text{Sinh}[d\,(a+b\,\text{Log}[x])]^p == \frac{(m+1)^p}{2^p\,b^p\,d^p\,p^p}\left(-e^{-\frac{a\,b\,d^2\,p}{m+1}}\,x^{-\frac{m+1}{p}}+e^{\frac{a\,b\,d^2\,p}{m+1}}\,x^{\frac{m+1}{p}}\right)^p$

Basis: If $b^2\,d^2\,p^2-(m+1)^2==0 \land p\in\mathbb{Z}$, then $\text{Cosh}[d\,(a+b\,\text{Log}[x])]^p == \frac{1}{2^p}\left(e^{-\frac{a\,b\,d^2\,p}{m+1}}\,x^{-\frac{m+1}{p}}+e^{\frac{a\,b\,d^2\,p}{m+1}}\,x^{\frac{m+1}{p}}\right)^p$

Note: The above identities need to be formally derived, and possibly the domain of p expanded.

Rule: If $p\in\mathbb{Z}^+ \land b^2\,d^2\,p^2-(m+1)^2==0$, then

$$\int (e\,x)^m \,\text{Sinh}[d\,(a+b\,\text{Log}[x])]^p\,dx \rightarrow \frac{(m+1)^p}{2^p\,b^p\,d^p\,p^p}\int \text{ExpandIntegrand}\left[(e\,x)^m\left(-e^{-\frac{a\,b\,d^2\,p}{m+1}}\,x^{-\frac{m+1}{p}}+e^{\frac{a\,b\,d^2\,p}{m+1}}\,x^{\frac{m+1}{p}}\right)^p,\,x\right]dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  (m+1)^p/(2^p*b^p*d^p*p^p)*
    Int[ExpandIntegrand[(e*x)^m*(-E^(-a*b*d^2*p/(m+1))*x^(-(m+1)/p)+E^(a*b*d^2*p/(m+1))*x^((m+1)/p))^p,x],x] /;
FreeQ[{a,b,d,e,m},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2-(m+1)^2,0]
```

```
Int[(e_.*x_)^m_.*Cosh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/2^p*Int[ExpandIntegrand[(e*x)^m*(E^(-a*b*d^2*p/(m+1))*x^(-(m+1)/p)+E^(a*b*d^2*p/(m+1))*x^((m+1)/p))^p,x],x] /;
FreeQ[{a,b,d,e,m},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2-(m+1)^2,0]
```

x: $\int (e\,x)^m \operatorname{Sinh}[d\,(a+b\operatorname{Log}[x])]^p\,dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\operatorname{Sinh}[d\,(a+b\operatorname{Log}[x])] == \frac{e^{a d}}{2} x^{b d} \left(1-e^{-2 a d} x^{-2 b d}\right)$

Basis: $\operatorname{Cosh}[d\,(a+b\operatorname{Log}[x])] == \frac{e^{a d}}{2} x^{b d} \left(1+e^{-2 a d} x^{-2 b d}\right)$

Rule: If $p \in \mathbb{Z}$, then

$$\int (e\,x)^m \operatorname{Sinh}[d\,(a+b\operatorname{Log}[x])]^p\,dx \to \frac{e^{a d p}}{2^p} \int (e\,x)^m x^{b d p} \left(1-e^{-2 a d} x^{-2 b d}\right)^p dx$$

Program code:

```
(* Int[(e_.*x_)^m_.*Sinh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  E^(a*d*p)/2^p*Int[(e*x)^m*x^(b*d*p)*(1-1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p] *)
```

```
(* Int[(e_.*x_)^m_.*Cosh[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  E^(a*d*p)/2^p*Int[(e*x)^m*x^(b*d*p)*(1+1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p] *)
```

2: $\int (e\,x)^m \operatorname{Sinh}[d\,(a+b\operatorname{Log}[x])]^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\operatorname{Sinh}[d\,(a+b\operatorname{Log}[x])]^p}{x^{b\,d\,p}\,\left(1-e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p} == 0$

Basis: $\partial_x \frac{\operatorname{Cosh}[d\,(a+b\operatorname{Log}[x])]^p}{x^{b\,d\,p}\,\left(1+e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int (e\,x)^m \operatorname{Sinh}[d\,(a+b\operatorname{Log}[x])]^p\,dx \longrightarrow \frac{\operatorname{Sinh}[d\,(a+b\operatorname{Log}[x])]^p}{x^{b\,d\,p}\,\left(1-e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p}\int (e\,x)^m\,x^{b\,d\,p}\,\left(1-e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Sinh[d*(a+b*Log[x])]^p/(x^(b*d*p)*(1-1/(E^(2*a*d)*x^(2*b*d)))^p)*
    Int[(e*x)^m*x^(b*d*p)*(1-1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

```
Int[(e_.*x_)^m_.*Cosh[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Cosh[d*(a+b*Log[x])]^p/(x^(b*d*p)*(1+1/(E^(2*a*d)*x^(2*b*d)))^p)*
    Int[(e*x)^m*x^(b*d*p)*(1+1/(E^(2*a*d)*x^(2*b*d)))^p,x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

3: $\int (e\,x)^m \,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p \,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\,\partial_x\,(c\,x^n)$

Rule:

$$\int (e\,x)^m \,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p \,dx \;\longrightarrow\; \frac{(e\,x)^{m+1}}{e\,(c\,x^n)^{(m+1)/n}} \int \frac{(c\,x^n)^{(m+1)/n}\,\text{Sinh}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p}{x}\,dx$$

$$\longrightarrow\; \frac{(e\,x)^{m+1}}{e\,n\,(c\,x^n)^{(m+1)/n}}\,\text{Subst}\left[\int x^{(m+1)/n-1}\,\text{Sinh}\left[d\,\left(a+b\,\text{Log}[x]\right)\right]^p\,dx,\,x,\,c\,x^n\right]$$

```
Int[(e_.*x_)^m_.*Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Sinh[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[(e_.*x_)^m_.*Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Cosh[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

3: $\int (h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\text{Sinh}[d\,(a+b\,\text{Log}[z])] == -\frac{1}{2}\,e^{-a\,d}\,z^{-b\,d}+\frac{1}{2}\,e^{a\,d}\,z^{b\,d}$

Basis: $\text{Cosh}[d\,(a+b\,\text{Log}[z])] == \frac{1}{2}\,e^{-a\,d}\,z^{-b\,d}+\frac{1}{2}\,e^{a\,d}\,z^{b\,d}$

Rule:

$$\int (h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx \to$$
$$-\frac{e^{-a\,d}\,(c\,x^n)^{-b\,d}}{2\,x^{-b\,d\,n}}\int x^{-b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx+\frac{e^{a\,d}\,(c\,x^n)^{b\,d}}{2\,x^{b\,d\,n}}\int x^{b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx$$

Program code:

```
Int[(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  -E^(-a*d)*(c*x^n)^(-b*d)/(2*x^(-b*d*n))*Int[x^(-b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] +
  E^(a*d)*(c*x^n)^(b*d)/(2*x^(b*d*n))*Int[x^(b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,q},x]
```

```
Int[(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  E^(-a*d)*(c*x^n)^(-b*d)/(2*x^(-b*d*n))*Int[x^(-b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] +
  E^(a*d)*(c*x^n)^(b*d)/(2*x^(b*d*n))*Int[x^(b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,q},x]
```

4: $\int (i\,x)^r\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\text{Sinh}[d\,(a+b\,\text{Log}[z])] == -\frac{1}{2}\,e^{-a\,d}\,z^{-b\,d}+\frac{1}{2}\,e^{a\,d}\,z^{b\,d}$

Basis: $\text{Cosh}[d\,(a+b\,\text{Log}[z])] == \frac{1}{2}\,e^{-a\,d}\,z^{-b\,d}+\frac{1}{2}\,e^{a\,d}\,z^{b\,d}$

Rule:

$$\int (i\,x)^r\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sinh}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx\,\longrightarrow$$
$$-\frac{e^{-a\,d}\,(i\,x)^r\,(c\,x^n)^{-b\,d}}{2\,x^{r-b\,d\,n}}\int x^{r-b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx+\frac{e^{a\,d}\,(i\,x)^r\,(c\,x^n)^{b\,d}}{2\,x^{r+b\,d\,n}}\int x^{r+b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx$$

Program code:

```
Int[(i_.*x_)^r_.*(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Sinh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  -E^(-a*d)*(i*x)^r*(c*x^n)^(-b*d)/(2*x^(r-b*d*n))*Int[x^(r-b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] +
  E^(a*d)*(i*x)^r*(c*x^n)^(b*d)/(2*x^(r+b*d*n))*Int[x^(r+b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,m,n,q,r},x]
```

```
Int[(i_.*x_)^r_.*(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Cosh[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  E^(-a*d)*(i*x)^r*(c*x^n)^(-b*d)/(2*x^(r-b*d*n))*Int[x^(r-b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] +
  E^(a*d)*(i*x)^r*(c*x^n)^(b*d)/(2*x^(r+b*d*n))*Int[x^(r+b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,m,n,q,r},x]
```

2. $\int u \operatorname{Sech}[d (a + b \operatorname{Log}[c x^n])]^p \, dx$

1. $\int \operatorname{Sech}[d (a + b \operatorname{Log}[c x^n])]^p \, dx$

1. $\int \operatorname{Sech}[d (a + b \operatorname{Log}[x])]^p \, dx$

1: $\int \operatorname{Sech}[d (a + b \operatorname{Log}[x])]^p \, dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\operatorname{Sech}[d (a + b \operatorname{Log}[x])] == \frac{2 e^{-a d} x^{-b d}}{1+e^{-2 a d} x^{-2 b d}}$

Basis: $\operatorname{Csch}[d (a + b \operatorname{Log}[x])] == \frac{2 e^{-a d} x^{-b d}}{1-e^{-2 a d} x^{-2 b d}}$

Rule: If $p \in \mathbb{Z}$, then

$$\int \operatorname{Sech}[d (a + b \operatorname{Log}[x])]^p \, dx \rightarrow 2^p e^{-a d p} \int \frac{x^{-b d p}}{\left(1 + e^{-2 a d} x^{-2 b d}\right)^p} \, dx$$

Program code:

```
Int[Sech[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  2^p*E^(-a*d*p)*Int[x^(-b*d*p)/(1+E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d},x] && IntegerQ[p]
```

```
Int[Csch[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  2^p*E^(-a*d*p)*Int[x^(-b*d*p)/(1-E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d},x] && IntegerQ[p]
```

2: $\int \text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\partial_x \frac{\text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,(1+e^{-2\,a\,d}\,x^{-2\,b\,d})^p}{x^{-b\,d\,p}} == 0$

Basis: $\partial_x \frac{\text{Csch}[d\,(a+b\,\text{Log}[x])]^p\,(1-e^{-2\,a\,d}\,x^{-2\,b\,d})^p}{x^{-b\,d\,p}} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int \text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,dx \longrightarrow \frac{\text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,(1+e^{-2\,a\,d}\,x^{-2\,b\,d})^p}{x^{-b\,d\,p}}\int \frac{x^{-b\,d\,p}}{(1+e^{-2\,a\,d}\,x^{-2\,b\,d})^p}\,dx$$

Program code:

```
Int[Sech[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Sech[d*(a+b*Log[x])]^p*(1+E^(-2*a*d)*x^(-2*b*d))^p/x^(-b*d*p)*
    Int[x^(-b*d*p)/(1+E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

```
Int[Csch[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Csch[d*(a+b*Log[x])]^p*(1-E^(-2*a*d)*x^(-2*b*d))^p/x^(-b*d*p)*
    Int[x^(-b*d*p)/(1-E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

2: $\int \text{Sech}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\partial_x\,(c\,x^n)$

Rule:

$$\int \text{Sech}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx \to \frac{x}{(c\,x^n)^{1/n}}\int \frac{(c\,x^n)^{1/n}\,\text{Sech}[d\,(a+b\,\text{Log}[c\,x^n])]^p}{x}\,dx$$

$$\to \frac{x}{n\,(c\,x^n)^{1/n}}\,\text{Subst}\left[\int x^{1/n-1}\,\text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,dx,\,x,\,c\,x^n\right]$$

```
Int[Sech[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Sech[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[Csch[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Csch[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

2. $\int (e\,x)^m\,\text{Sech}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

   1. $\int (e\,x)^m\,\text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,dx$

      1: $\int (e\,x)^m\,\text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Sech}[d\,(a+b\,\text{Log}[x])]==\frac{2\,e^{-a\,d}\,x^{-b\,d}}{1+e^{-2\,a\,d}\,x^{-2\,b\,d}}$

Basis: $\text{Csch}[d\,(a+b\,\text{Log}[x])]==\frac{2\,e^{-a\,d}\,x^{-b\,d}}{1-e^{-2\,a\,d}\,x^{-2\,b\,d}}$

Rule: If $p\in\mathbb{Z}$, then

$$\int (e\,x)^m\,\text{Sech}[d\,(a+b\,\text{Log}[x])]^p\,dx\;\longrightarrow\;2^p\,e^{-a\,d\,p}\int\frac{(e\,x)^m\,x^{-b\,d\,p}}{\left(1+e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sech[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  2^p*E^(-a*d*p)*Int[(e*x)^m*x^(-b*d*p)/(1+E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p]
```

```
Int[(e_.*x_)^m_.*Csch[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  2^p*E^(-a*d*p)*Int[(e*x)^m*x^(-b*d*p)/(1-E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p]
```

2: $\int (e\,x)^m \operatorname{Sech}[d\,(a+b\operatorname{Log}[x])]^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\partial_x \frac{\operatorname{Sech}[d\,(a+b\operatorname{Log}[x])]^p\,\left(1+e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p}{x^{-b\,d\,p}} == 0$

Basis: $\partial_x \frac{\operatorname{Csch}[d\,(a+b\operatorname{Log}[x])]^p\,\left(1-e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p}{x^{-b\,d\,p}} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int (e\,x)^m \operatorname{Sech}[d\,(a+b\operatorname{Log}[x])]^p\,dx \rightarrow \frac{\operatorname{Sech}[d\,(a+b\operatorname{Log}[x])]^p\,\left(1+e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p}{x^{-b\,d\,p}}\int \frac{(e\,x)^m\,x^{-b\,d\,p}}{\left(1+e^{-2\,a\,d}\,x^{-2\,b\,d}\right)^p}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sech[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Sech[d*(a+b*Log[x])]^p*(1+E^(-2*a*d)*x^(-2*b*d))^p/x^(-b*d*p)*
    Int[(e*x)^m*x^(-b*d*p)/(1+E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

```
Int[(e_.*x_)^m_.*Csch[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Csch[d*(a+b*Log[x])]^p*(1-E^(-2*a*d)*x^(-2*b*d))^p/x^(-b*d*p)*
    Int[(e*x)^m*x^(-b*d*p)/(1-E^(-2*a*d)*x^(-2*b*d))^p,x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

2: $\int (e\,x)^m\,\text{Sech}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\,\partial_x\,(c\,x^n)$

Rule:

$$\int (e\,x)^m\,\text{Sech}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p\,dx \,\to\, \frac{(e\,x)^{m+1}}{e\,\left(c\,x^n\right)^{(m+1)/n}}\int \frac{\left(c\,x^n\right)^{(m+1)/n}\,\text{Sech}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p}{x}\,dx$$

$$\to\,\frac{(e\,x)^{m+1}}{e\,n\,\left(c\,x^n\right)^{(m+1)/n}}\,\text{Subst}\left[\int x^{(m+1)/n-1}\,\text{Sech}\left[d\,\left(a+b\,\text{Log}[x]\right)\right]^p\,dx,\,x,\,c\,x^n\right]$$

```
Int[(e_.*x_)^m_.*Sech[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Sech[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[(e_.*x_)^m_.*Csch[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Csch[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

3. $\int u \operatorname{Sinh}[a\, x^n \operatorname{Log}[b\, x]] \operatorname{Log}[b\, x] \, dx$

1: $\int \operatorname{Sinh}[a\, x \operatorname{Log}[b\, x]] \operatorname{Log}[b\, x] \, dx$

Rule:

$$\int \operatorname{Sinh}[a\, x \operatorname{Log}[b\, x]] \operatorname{Log}[b\, x] \, dx \longrightarrow \frac{\operatorname{Cosh}[a\, x \operatorname{Log}[b\, x]]}{a} - \int \operatorname{Sinh}[a\, x \operatorname{Log}[b\, x]] \, dx$$

Program code:

```
Int[Sinh[a_.*x_*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  Cosh[a*x*Log[b*x]]/a - Int[Sinh[a*x*Log[b*x]],x] /;
FreeQ[{a,b},x]
```

```
Int[Cosh[a_.*x_*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  Sinh[a*x*Log[b*x]]/a - Int[Cosh[a*x*Log[b*x]],x] /;
FreeQ[{a,b},x]
```

2: $\int x^m \operatorname{Sinh}[a\, x^n \operatorname{Log}[b\, x]] \operatorname{Log}[b\, x] \, dx$ when $m == n - 1$

Rule: If $m == n - 1$, then

$$\int x^m \operatorname{Sinh}[a\, x^n \operatorname{Log}[b\, x]] \operatorname{Log}[b\, x] \, dx \longrightarrow \frac{\operatorname{Cosh}[a\, x^n \operatorname{Log}[b\, x]]}{a\, n} - \frac{1}{n} \int x^m \operatorname{Sinh}[a\, x^n \operatorname{Log}[b\, x]] \, dx$$

Program code:

```
Int[x_^m_.*Sinh[a_.*x_^n_.*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  Cosh[a*x^n*Log[b*x]]/(a*n) - 1/n*Int[x^m*Sinh[a*x^n*Log[b*x]],x] /;
FreeQ[{a,b,m,n},x] && EqQ[m,n-1]
```

```
Int[x_^m_.*Cosh[a_.*x_^n_.*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  Sinh[a*x^n*Log[b*x]]/(a*n) - 1/n*Int[x^m*Cosh[a*x^n*Log[b*x]],x] /;
FreeQ[{a,b,m,n},x] && EqQ[m,n-1]
```