# Rules for integrands of the form $u\,(e+f\,x)^m\,(a+b\,\mathrm{Trig}[c+d\,x])^p$

1. $\int\frac{(e+f\,x)^m\,\mathrm{Trig}[c+d\,x]^n}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$

   1: $\int\frac{(e+f\,x)^m\,\mathrm{Sin}[c+d\,x]^n}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$ when $(m\mid n)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{z^n}{a+b\,z}==\frac{z^{n-1}}{b}-\frac{a\,z^{n-1}}{b\,(a+b\,z)}$

Rule: If $(m\mid n)\in\mathbb{Z}^+$, then

$$\int\frac{(e+f\,x)^m\,\mathrm{Sin}[c+d\,x]^n}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx\;\to\;\frac{1}{b}\int(e+f\,x)^m\,\mathrm{Sin}[c+d\,x]^{n-1}\,dx-\frac{a}{b}\int\frac{(e+f\,x)^m\,\mathrm{Sin}[c+d\,x]^{n-1}}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sin[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Sin[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Cos[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Cos[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

2. $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $n \in \mathbb{Z}^+$

1. $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+$

1: $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+ \wedge a^2-b^2 == 0$

Derivation: Algebraic expansion

Basis: If $a^2-b^2 == 0$, then $\frac{\operatorname{Cos}[z]}{a+b \operatorname{Sin}[z]} == \frac{i}{b}+\frac{2}{i b+a e^{i z}} == -\frac{i}{b}+\frac{2 e^{i z}}{a-i b e^{i z}}$

Basis: If $a^2-b^2 == 0$, then $\frac{\operatorname{Sin}[z]}{a+b \operatorname{Cos}[z]} == -\frac{i}{b}+\frac{2 i}{b+a e^{i z}} == \frac{i}{b}-\frac{2 i e^{i z}}{a+b e^{i z}}$

Note: Although the first expansion is simpler, the second is used so the antiderivative will be expressed in terms of $e^{i (c+d x)}$ rather than $e^{-i (c+d x)}$.

Rule: If $m \in \mathbb{Z}^+ \wedge a^2-b^2 == 0$, then

$$\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]}{a+b \operatorname{Sin}[c+d x]} \, dx \rightarrow -\frac{i (e+f x)^{m+1}}{b f (m+1)}+2 \int \frac{(e+f x)^m e^{i (c+d x)}}{a-i b e^{i (c+d x)}} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]/(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  -I*(e+f*x)^(m+1)/(b*f*(m+1)) + 2*Int[(e+f*x)^m*E^(I*(c+d*x))/(a-I*b*E^(I*(c+d*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[a^2-b^2,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]/(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  I*(e+f*x)^(m+1)/(b*f*(m+1)) - 2*I*Int[(e+f*x)^m*E^(I*(c+d*x))/(a+b*E^(I*(c+d*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && EqQ[a^2-b^2,0]
```

**2:** $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+ \wedge a^2-b^2>0$

## Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Cos}[z]}{a+b \operatorname{Sin}[z]} == \frac{i}{b} + \frac{1}{i b+\left(a-\sqrt{a^2-b^2}\right) e^{i z}} + \frac{1}{i b+\left(a+\sqrt{a^2-b^2}\right) e^{i z}} == -\frac{i}{b} + \frac{e^{i z}}{a-\sqrt{a^2-b^2}-i b e^{i z}} + \frac{e^{i z}}{a+\sqrt{a^2-b^2}-i b e^{i z}}$

Basis: $\frac{\operatorname{Sin}[z]}{a+b \operatorname{Cos}[z]} == -\frac{i}{b} + \frac{i}{b+\left(a-\sqrt{a^2-b^2}\right) e^{i z}} + \frac{i}{b+\left(a+\sqrt{a^2-b^2}\right) e^{i z}} == \frac{i}{b} - \frac{i e^{i z}}{a-\sqrt{a^2-b^2}+b e^{i z}} - \frac{i e^{i z}}{a+\sqrt{a^2-b^2}+b e^{i z}}$

Note: Although the first expansion is simpler, the second is used so the antiderivative will be expressed in terms of $e^{i (c+d x)}$ rather than $e^{-i (c+d x)}$.

Rule: If $m \in \mathbb{Z}^+ \wedge a^2-b^2>0$, then

$$\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]}{a+b \operatorname{Sin}[c+d x]} \, dx \rightarrow -\frac{i (e+f x)^{m+1}}{b f (m+1)} + \int \frac{(e+f x)^m e^{i (c+d x)}}{a-\sqrt{a^2-b^2}-i b e^{i (c+d x)}} \, dx + \int \frac{(e+f x)^m e^{i (c+d x)}}{a+\sqrt{a^2-b^2}-i b e^{i (c+d x)}} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]/(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  -I*(e+f*x)^(m+1)/(b*f*(m+1)) +
  Int[(e+f*x)^m*E^(I*(c+d*x))/(a-Rt[a^2-b^2,2]-I*b*E^(I*(c+d*x))),x] +
  Int[(e+f*x)^m*E^(I*(c+d*x))/(a+Rt[a^2-b^2,2]-I*b*E^(I*(c+d*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && PosQ[a^2-b^2]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]/(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  I*(e+f*x)^(m+1)/(b*f*(m+1)) -
  I*Int[(e+f*x)^m*E^(I*(c+d*x))/(a-Rt[a^2-b^2,2]+b*E^(I*(c+d*x))),x] -
  I*Int[(e+f*x)^m*E^(I*(c+d*x))/(a+Rt[a^2-b^2,2]+b*E^(I*(c+d*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && PosQ[a^2-b^2]
```

**3:** $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+ \wedge a^2-b^2 \ngeq 0$

## Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Cos}[z]}{a+b \operatorname{Sin}[z]} = -\frac{i}{b}+\frac{i e^{i z}}{i a-\sqrt{-a^2+b^2}+b e^{i z}}+\frac{i e^{i z}}{i a+\sqrt{-a^2+b^2}+b e^{i z}}$

Basis: $\frac{\operatorname{Sin}[z]}{a+b \operatorname{Cos}[z]} = \frac{i}{b}+\frac{e^{i z}}{i a-\sqrt{-a^2+b^2}+i b e^{i z}}+\frac{e^{i z}}{i a+\sqrt{-a^2+b^2}+i b e^{i z}}$

Rule: If $m \in \mathbb{Z}^+ \wedge a^2-b^2 \ngeq 0$, then

$$\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]}{a+b \operatorname{Sin}[c+d x]} \, dx \longrightarrow -\frac{i (e+f x)^{m+1}}{b f (m+1)}+i \int \frac{(e+f x)^m e^{i (c+d x)}}{i a-\sqrt{-a^2+b^2}+b e^{i (c+d x)}} \, dx+i \int \frac{(e+f x)^m e^{i (c+d x)}}{i a+\sqrt{-a^2+b^2}+b e^{i (c+d x)}} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]/(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  -I*(e+f*x)^(m+1)/(b*f*(m+1)) +
  I*Int[(e+f*x)^m*E^(I*(c+d*x))/(I*a-Rt[-a^2+b^2,2]+b*E^(I*(c+d*x))),x] +
  I*Int[(e+f*x)^m*E^(I*(c+d*x))/(I*a+Rt[-a^2+b^2,2]+b*E^(I*(c+d*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NegQ[a^2-b^2]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]/(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  I*(e+f*x)^(m+1)/(b*f*(m+1)) +
  Int[(e+f*x)^m*E^(I*(c+d*x))/(I*a-Rt[-a^2+b^2,2]+I*b*E^(I*(c+d*x))),x] +
  Int[(e+f*x)^m*E^(I*(c+d*x))/(I*a+Rt[-a^2+b^2,2]+I*b*E^(I*(c+d*x))),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NegQ[a^2-b^2]
```

2. $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $n-1 \in \mathbb{Z}^+$

**1:** $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $n-1 \in \mathbb{Z}^+ \land a^2-b^2 == 0$

## Derivation: Algebraic expansion

Basis: If $a^2-b^2 == 0$, then $\frac{\operatorname{Cos}[z]^2}{a+b \operatorname{Sin}[z]} == \frac{1}{a} - \frac{\operatorname{Sin}[z]}{b}$

Rule: If $n-1 \in \mathbb{Z}^+ \land a^2-b^2 == 0$, then

$$\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx \to \frac{1}{a} \int (e+f x)^m \operatorname{Cos}[c+d x]^{n-2} \, dx - \frac{1}{b} \int (e+f x)^m \operatorname{Cos}[c+d x]^{n-2} \operatorname{Sin}[c+d x] \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^n_/(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Cos[c+d*x]^(n-2),x] -
  1/b*Int[(e+f*x)^m*Cos[c+d*x]^(n-2)*Sin[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[n,1] && EqQ[a^2-b^2,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^n_/(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sin[c+d*x]^(n-2),x] -
  1/b*Int[(e+f*x)^m*Sin[c+d*x]^(n-2)*Cos[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[n,1] && EqQ[a^2-b^2,0]
```

**2:** $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $n-1 \in \mathbb{Z}^+ \land a^2-b^2 \neq 0 \land m \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Cos}[z]^2}{a+b \operatorname{Sin}[z]} == \frac{a}{b^2} - \frac{\operatorname{Sin}[z]}{b} - \frac{a^2-b^2}{b^2 (a+b \operatorname{Sin}[z])}$

Basis: $\frac{\operatorname{Sin}[z]^2}{a+b \operatorname{Cos}[z]} == \frac{a}{b^2} - \frac{\operatorname{Cos}[z]}{b} - \frac{a^2-b^2}{b^2 (a+b \operatorname{Cos}[z])}$

Rule: If $n-1 \in \mathbb{Z}^+ \land a^2-b^2 \neq 0 \land m \in \mathbb{Z}^+$, then

$$\int \frac{(e+f\,x)^m \operatorname{Cos}[c+d\,x]^n}{a+b\operatorname{Sin}[c+d\,x]}\,dx \longrightarrow$$

$$\frac{a}{b^2}\int (e+f\,x)^m \operatorname{Cos}[c+d\,x]^{n-2}\,dx-\frac{1}{b}\int (e+f\,x)^m \operatorname{Cos}[c+d\,x]^{n-2}\operatorname{Sin}[c+d\,x]\,dx-\frac{a^2-b^2}{b^2}\int \frac{(e+f\,x)^m \operatorname{Cos}[c+d\,x]^{n-2}}{a+b\operatorname{Sin}[c+d\,x]}\,dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^n_/(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  a/b^2*Int[(e+f*x)^m*Cos[c+d*x]^(n-2),x] -
  1/b*Int[(e+f*x)^m*Cos[c+d*x]^(n-2)*Sin[c+d*x],x] -
  (a^2-b^2)/b^2*Int[(e+f*x)^m*Cos[c+d*x]^(n-2)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[n,1] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^n_/(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  a/b^2*Int[(e+f*x)^m*Sin[c+d*x]^(n-2),x] -
  1/b*Int[(e+f*x)^m*Sin[c+d*x]^(n-2)*Cos[c+d*x],x] -
  (a^2-b^2)/b^2*Int[(e+f*x)^m*Sin[c+d*x]^(n-2)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[n,1] && NeQ[a^2-b^2,0] && IGtQ[m,0]
```

**3:** $\int \frac{(e+f x)^m \operatorname{Tan}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Tan}[z]^p}{a+b \operatorname{Sin}[z]} == \frac{\operatorname{Sec}[z] \operatorname{Tan}[z]^{p-1}}{b} - \frac{a \operatorname{Sec}[z] \operatorname{Tan}[z]^{p-1}}{b (a+b \operatorname{Sin}[z])}$

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Tan}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx \to \frac{1}{b} \int (e+f x)^m \operatorname{Sec}[c+d x] \operatorname{Tan}[c+d x]^{n-1} \, dx - \frac{a}{b} \int \frac{(e+f x)^m \operatorname{Sec}[c+d x] \operatorname{Tan}[c+d x]^{n-1}}{a+b \operatorname{Sin}[c+d x]} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Tan[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sec[c+d*x]*Tan[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Sec[c+d*x]*Tan[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Cot[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Csc[c+d*x]*Cot[c+d*x]^(n-1),x] - a/b*Int[(e+f*x)^m*Csc[c+d*x]*Cot[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

**4:** $\int \frac{(e+f x)^m \operatorname{Cot}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Cot}[z]^n}{a+b \operatorname{Sin}[z]} == \frac{\operatorname{Cot}[z]^n}{a} - \frac{b \operatorname{Cos}[z] \operatorname{Cot}[z]^{n-1}}{a (a+b \operatorname{Sin}[z])}$

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Cot}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} dx \to \frac{1}{a} \int (e+f x)^m \operatorname{Cot}[c+d x]^n dx - \frac{b}{a} \int \frac{(e+f x)^m \operatorname{Cos}[c+d x] \operatorname{Cot}[c+d x]^{n-1}}{a+b \operatorname{Sin}[c+d x]} dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cot[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Cot[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Cos[c+d*x]*Cot[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Tan[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Tan[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Sin[c+d*x]*Tan[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

5. $\int \frac{(e+f x)^m \operatorname{Sec}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} dx$

1: $\int \frac{(e+f x)^m \operatorname{Sec}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} dx$ when $m \in \mathbb{Z}^+ \land a^2 - b^2 == 0$

## Derivation: Algebraic expansion

## Basis: If $a^2 - b^2 == 0$, then $\frac{1}{a+b \operatorname{Sin}[z]} == \frac{\operatorname{Sec}[z]^2}{a} - \frac{\operatorname{Sec}[z] \operatorname{Tan}[z]}{b}$

## Rule: If $m \in \mathbb{Z}^+ \land a^2 - b^2 == 0$, then

$$\int \frac{(e+f x)^m \operatorname{Sec}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} dx \to \frac{1}{a} \int (e+f x)^m \operatorname{Sec}[c+d x]^{n+2} dx - \frac{1}{b} \int (e+f x)^m \operatorname{Sec}[c+d x]^{n+1} \operatorname{Tan}[c+d x] dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Sec[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sec[c+d*x]^(n+2),x] -
  1/b*Int[(e+f*x)^m*Sec[c+d*x]^(n+1)*Tan[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && EqQ[a^2-b^2,0]
```

```
Int[(e_.+f_.*x_)^m_.*Csc[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Csc[c+d*x]^(n+2),x] -
  1/b*Int[(e+f*x)^m*Csc[c+d*x]^(n+1)*Cot[c+d*x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && EqQ[a^2-b^2,0]
```

**2:** $\int \frac{(e+f x)^m \operatorname{Sec}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $m \in \mathbb{Z}^+ \land a^2-b^2 \neq 0 \land n \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Sec}[z]^2}{a+b \operatorname{Sin}[z]} = -\frac{b^2}{\left(a^2-b^2\right) (a+b \operatorname{Sin}[z])}+\frac{\operatorname{Sec}[z]^2 (a-b \operatorname{Sin}[z])}{a^2-b^2}$

Basis: $\frac{\operatorname{Csc}[z]^2}{a+b \operatorname{Cos}[z]} = -\frac{b^2}{\left(a^2-b^2\right) (a+b \operatorname{Cos}[z])}+\frac{\operatorname{Csc}[z]^2 (a-b \operatorname{Cos}[z])}{a^2-b^2}$

Rule: If $m \in \mathbb{Z}^+ \land a^2-b^2 \neq 0 \land n \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Sec}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx \to -\frac{b^2}{a^2-b^2} \int \frac{(e+f x)^m \operatorname{Sec}[c+d x]^{n-2}}{a+b \operatorname{Sin}[c+d x]} \, dx+\frac{1}{a^2-b^2} \int (e+f x)^m \operatorname{Sec}[c+d x]^n (a-b \operatorname{Sin}[c+d x]) \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Sec[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  -b^2/(a^2-b^2)*Int[(e+f*x)^m*Sec[c+d*x]^(n-2)/(a+b*Sin[c+d*x]),x] +
  1/(a^2-b^2)*Int[(e+f*x)^m*Sec[c+d*x]^n*(a-b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[a^2-b^2,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Csc[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  -b^2/(a^2-b^2)*Int[(e+f*x)^m*Csc[c+d*x]^(n-2)/(a+b*Cos[c+d*x]),x] +
  1/(a^2-b^2)*Int[(e+f*x)^m*Csc[c+d*x]^n*(a-b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && NeQ[a^2-b^2,0] && IGtQ[n,0]
```

**6:** $\int \frac{(e+f x)^m \operatorname{Csc}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

Basis: $\frac{\operatorname{Csc}[z]^n}{a+b \operatorname{Sin}[z]} = \frac{\operatorname{Csc}[z]^n}{a} - \frac{b \operatorname{Csc}[z]^{n-1}}{a (a+b \operatorname{Sin}[z])}$

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f x)^m \operatorname{Csc}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx \longrightarrow \frac{1}{a} \int (e+f x)^m \operatorname{Csc}[c+d x]^n \, dx - \frac{b}{a} \int \frac{(e+f x)^m \operatorname{Csc}[c+d x]^{n-1}}{a+b \operatorname{Sin}[c+d x]} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Csc[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Csc[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Csc[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sec[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sec[c+d*x]^n,x] - b/a*Int[(e+f*x)^m*Sec[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

**U:** $\int \frac{(e+f\,x)^m\,\text{Trig}[c+d\,x]^n}{a+b\,\text{Sin}[c+d\,x]}\,dx$

## Rule:

$$\int \frac{(e+f\,x)^m\,\text{Trig}[c+d\,x]^n}{a+b\,\text{Sin}[c+d\,x]}\,dx \,\longrightarrow\, \int \frac{(e+f\,x)^m\,\text{Trig}[c+d\,x]^n}{a+b\,\text{Sin}[c+d\,x]}\,dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*F[c+d*x]^n/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && TrigQ[F]
```

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*F[c+d*x]^n/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && TrigQ[F]
```

2. $\int \frac{(e+f\,x)^m \operatorname{Cos}[c+d\,x]^p \operatorname{Trig}[c+d\,x]^n}{a+b\operatorname{Sin}[c+d\,x]}\,dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

1: $\int \frac{(e+f\,x)^m \operatorname{Cos}[c+d\,x]^p \operatorname{Sin}[c+d\,x]^n}{a+b\operatorname{Sin}[c+d\,x]}\,dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{z^n}{a+b\,z} == \frac{z^{n-1}}{b} - \frac{a\,z^{n-1}}{b\,(a+b\,z)}$

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+f\,x)^m \operatorname{Cos}[c+d\,x]^p \operatorname{Sin}[c+d\,x]^n}{a+b\operatorname{Sin}[c+d\,x]}\,dx \to \frac{1}{b}\int (e+f\,x)^m \operatorname{Cos}[c+d\,x]^p \operatorname{Sin}[c+d\,x]^{n-1}\,dx - \frac{a}{b}\int \frac{(e+f\,x)^m \operatorname{Cos}[c+d\,x]^p \operatorname{Sin}[c+d\,x]^{n-1}}{a+b\operatorname{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^p_.*Sin[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Cos[c+d*x]^p*Sin[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Cos[c+d*x]^p*Sin[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^p_.*Cos[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sin[c+d*x]^p*Cos[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Sin[c+d*x]^p*Cos[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

**2:** $\int \frac{(e+f\,x)^m \,\mathrm{Cos}[c+d\,x]^p \,\mathrm{Tan}[c+d\,x]^n}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$ when $(m\mid n\mid p)\in\mathbb{Z}^+$

## Derivation: Algebraic expansion

Basis: $\frac{\mathrm{Tan}[z]^p}{a+b\,\mathrm{Sin}[z]} = \frac{\mathrm{Sec}[z]\,\mathrm{Tan}[z]^{p-1}}{b} - \frac{a\,\mathrm{Sec}[z]\,\mathrm{Tan}[z]^{p-1}}{b\,(a+b\,\mathrm{Sin}[z])}$

Rule: If $(m\mid n\mid p)\in\mathbb{Z}^+$, then

$$\int \frac{(e+f\,x)^m \,\mathrm{Cos}[c+d\,x]^p \,\mathrm{Tan}[c+d\,x]^n}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx \longrightarrow \frac{1}{b}\int (e+f\,x)^m \,\mathrm{Cos}[c+d\,x]^{p-1}\,\mathrm{Tan}[c+d\,x]^{n-1}\,dx - \frac{a}{b}\int \frac{(e+f\,x)^m \,\mathrm{Cos}[c+d\,x]^{p-1}\,\mathrm{Tan}[c+d\,x]^{n-1}}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^p_.*Tan[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Cos[c+d*x]^(p-1)*Tan[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Cos[c+d*x]^(p-1)*Tan[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^p_.*Cot[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/b*Int[(e+f*x)^m*Sin[c+d*x]^(p-1)*Cot[c+d*x]^(n-1),x] -
  a/b*Int[(e+f*x)^m*Sin[c+d*x]^(p-1)*Cot[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

**3:** $\int \frac{(e+f\,x)^m \,\mathrm{Cos}[c+d\,x]^p \,\mathrm{Cot}[c+d\,x]^n}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$ when $(m\mid n\mid p)\in\mathbb{Z}^+$

## Derivation: Algebraic expansion

Basis: $\frac{\mathrm{Cot}[z]^n}{a+b\,\mathrm{Sin}[z]} = \frac{\mathrm{Cot}[z]^n}{a} - \frac{b\,\mathrm{Cot}[z]^{n-1}\,\mathrm{Cos}[z]}{a\,(a+b\,\mathrm{Sin}[z])}$

Rule: If $(m\mid n\mid p)\in\mathbb{Z}^+$, then

$$\int \frac{(e+fx)^m \operatorname{Cos}[c+dx]^p \operatorname{Cot}[c+dx]^n}{a+b\operatorname{Sin}[c+dx]} \, dx \to \frac{1}{a}\int (e+fx)^m \operatorname{Cos}[c+dx]^p \operatorname{Cot}[c+dx]^n \, dx - \frac{b}{a}\int \frac{(e+fx)^m \operatorname{Cos}[c+dx]^{p+1} \operatorname{Cot}[c+dx]^{n-1}}{a+b\operatorname{Sin}[c+dx]} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^p_.*Cot[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Cos[c+d*x]^p*Cot[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Cos[c+d*x]^(p+1)*Cot[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^p_.*Tan[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sin[c+d*x]^p*Tan[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Sin[c+d*x]^(p+1)*Tan[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

4: $\int \frac{(e+fx)^m \operatorname{Cos}[c+dx]^p \operatorname{Csc}[c+dx]^n}{a+b\operatorname{Sin}[c+dx]} \, dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

## Basis: $\frac{\operatorname{Csc}[z]^n}{a+b\operatorname{Sin}[z]} == \frac{\operatorname{Csc}[z]^n}{a} - \frac{b\operatorname{Csc}[z]^{n-1}}{a\,(a+b\operatorname{Sin}[z])}$

## Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int \frac{(e+fx)^m \operatorname{Cos}[c+dx]^p \operatorname{Csc}[c+dx]^n}{a+b\operatorname{Sin}[c+dx]} \, dx \to \frac{1}{a}\int (e+fx)^m \operatorname{Cos}[c+dx]^p \operatorname{Csc}[c+dx]^n \, dx - \frac{b}{a}\int \frac{(e+fx)^m \operatorname{Cos}[c+dx]^p \operatorname{Csc}[c+dx]^{n-1}}{a+b\operatorname{Sin}[c+dx]} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^p_.*Csc[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Cos[c+d*x]^p*Csc[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Cos[c+d*x]^p*Csc[c+d*x]^(n-1)/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^p_.*Sec[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[(e+f*x)^m*Sin[c+d*x]^p*Sec[c+d*x]^n,x] -
  b/a*Int[(e+f*x)^m*Sin[c+d*x]^p*Sec[c+d*x]^(n-1)/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

**U:** $\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^p \operatorname{Trig}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$

## Rule:

$$\int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^p \operatorname{Trig}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx \longrightarrow \int \frac{(e+f x)^m \operatorname{Cos}[c+d x]^p \operatorname{Trig}[c+d x]^n}{a+b \operatorname{Sin}[c+d x]} \, dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*Cos[c_.+d_.*x_]^p_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*Cos[c+d*x]^p*F[c+d*x]^n/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && TrigQ[F]
```

```
Int[(e_.+f_.*x_)^m_.*Sin[c_.+d_.*x_]^p_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  Unintegrable[(e+f*x)^m*Sin[c+d*x]^p*F[c+d*x]^n/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && TrigQ[F]
```

**3:** $\int \frac{(e+f x)^m \operatorname{Trig}[c+d x]^n}{a+b \operatorname{Sec}[c+d x]} \, dx$ when $(m \mid n) \in \mathbb{Z}$

## Derivation: Algebraic normalization

## Basis: $\frac{1}{a+b \operatorname{Sec}[z]} = \frac{\operatorname{Cos}[z]}{b+a \operatorname{Cos}[z]}$

## Rule: If $(m \mid n) \in \mathbb{Z}$, then

$$\int \frac{(e+f\,x)^m\,\text{Trig}[c+d\,x]^n}{a+b\,\text{Sec}[c+d\,x]}\,dx \;\to\; \int \frac{(e+f\,x)^m\,\text{Cos}[c+d\,x]\,\text{Trig}[c+d\,x]^n}{b+a\,\text{Cos}[c+d\,x]}\,dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Sec[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Cos[c+d*x]*F[c+d*x]^n/(b+a*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && TrigQ[F] && IntegersQ[m,n]
```

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_./(a_+b_.*Csc[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Sin[c+d*x]*F[c+d*x]^n/(b+a*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && TrigQ[F] && IntegersQ[m,n]
```

**4:** $\int \frac{(e+f\,x)^m\,\text{Trig1}[c+d\,x]^n\,\text{Trig2}[c+d\,x]^p}{a+b\,\text{Sec}[c+d\,x]}\,dx$ when $(m\mid n\mid p)\in\mathbb{Z}$

## Derivation: Algebraic normalization

## Basis: $\frac{1}{a+b\,\text{Sec}[z]} == \frac{\text{Cos}[z]}{b+a\,\text{Cos}[z]}$

## Rule: If $(m\mid n\mid p)\in\mathbb{Z}$, then

$$\int \frac{(e+f\,x)^m\,\text{Trig1}[c+d\,x]^n\,\text{Trig2}[c+d\,x]^p}{a+b\,\text{Sec}[c+d\,x]}\,dx \;\to\; \int \frac{(e+f\,x)^m\,\text{Cos}[c+d\,x]\,\text{Trig1}[c+d\,x]^n\,\text{Trig2}[c+d\,x]^p}{b+a\,\text{Cos}[c+d\,x]}\,dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_.*G_[c_.+d_.*x_]^p_./(a_+b_.*Sec[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Cos[c+d*x]*F[c+d*x]^n*G[c+d*x]^p/(b+a*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && TrigQ[F] && TrigQ[G] && IntegersQ[m,n,p]
```

```
Int[(e_.+f_.*x_)^m_.*F_[c_.+d_.*x_]^n_.*G_[c_.+d_.*x_]^p_./(a_+b_.*Csc[c_.+d_.*x_]),x_Symbol] :=
  Int[(e+f*x)^m*Sin[c+d*x]*F[c+d*x]^n*G[c+d*x]^p/(b+a*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && TrigQ[F] && TrigQ[G] && IntegersQ[m,n,p]
```

## Rules for integrands involving trig functions

**0.** $\int \text{Sin}[a+b\,x]^p\,\text{Trig}[c+d\,x]^q\,dx$

**1:** $\int \text{Sin}[a+b\,x]^p\,\text{Sin}[c+d\,x]^q\,dx$ when $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$

### Derivation: Algebraic expansion

Basis: $\text{Sin}[v]^p\,\text{Sin}[w]^q == \frac{1}{2^{p+q}}\left(i\,e^{-i\,v}-i\,e^{i\,v}\right)^p\left(i\,e^{-i\,w}-i\,e^{i\,w}\right)^q$

Rule: If $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$, then

$$\int \text{Sin}[a+b\,x]^p\,\text{Sin}[c+d\,x]^q\,dx \longrightarrow \frac{1}{2^{p+q}}\int\left(i\,e^{-i\,(c+d\,x)}-i\,e^{i\,(c+d\,x)}\right)^q\,\text{ExpandIntegrand}\left[\left(i\,e^{-i\,(a+b\,x)}-i\,e^{i\,(a+b\,x)}\right)^p,\,x\right]dx$$

### Program code:

```
Int[Sin[a_.+b_.*x_]^p_.*Sin[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(I/E^(I*(c+d*x))-I*E^(I*(c+d*x)))^q,(I/E^(I*(a+b*x))-I*E^(I*(a+b*x)))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

```
Int[Cos[a_.+b_.*x_]^p_.*Cos[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(E^(-I*(c+d*x))+E^(I*(c+d*x)))^q,(E^(-I*(a+b*x))+E^(I*(a+b*x)))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

**2:** $\int \mathrm{Sin}[a+b\,x]^p\,\mathrm{Cos}[c+d\,x]^q\,dx$ when $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$

### Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[v]^p\,\mathrm{Cos}[w]^q == \frac{1}{2^{p+q}}\left(i\,e^{-i\,v}-i\,e^{i\,v}\right)^p\left(e^{-i\,w}+e^{i\,w}\right)^q$

Rule: If $p\in\mathbb{Z}^+\land q\in\mathbb{Z}$, then

$$\int \mathrm{Sin}[a+b\,x]^p\,\mathrm{Cos}[c+d\,x]^q\,dx \to \frac{1}{2^{p+q}}\int\left(e^{-i\,(c+d\,x)}+e^{i\,(c+d\,x)}\right)^q\,\mathrm{ExpandIntegrand}\left[\left(i\,e^{-i\,(a+b\,x)}-i\,e^{i\,(a+b\,x)}\right)^p,\,x\right]dx$$

### Program code:

```
Int[Sin[a_.+b_.*x_]^p_.*Cos[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(E^(-I*(c+d*x))+E^(I*(c+d*x)))^q,(I/E^(I*(a+b*x))-I*E^(I*(a+b*x)))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

```
Int[Cos[a_.+b_.*x_]^p_.*Sin[c_.+d_.*x_]^q_.,x_Symbol] :=
  1/2^(p+q)*Int[ExpandIntegrand[(I/E^(I*(c+d*x))-I*E^(I*(c+d*x)))^q,(E^(-I*(a+b*x))+E^(I*(a+b*x)))^p,x],x] /;
FreeQ[{a,b,c,d,q},x] && IGtQ[p,0] && Not[IntegerQ[q]]
```

**3:** $\int \mathrm{Sin}[a+b\,x]\,\mathrm{Tan}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

### Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[v]\,\mathrm{Tan}[w] == \frac{e^{-i\,v}}{2}-\frac{e^{i\,v}}{2}-\frac{e^{-i\,v}}{1+e^{2\,i\,w}}+\frac{e^{i\,v}}{1+e^{2\,i\,w}}$

Basis: $\mathrm{Cos}[v]\,\mathrm{Cot}[w] == \frac{i\,e^{-i\,v}}{2}+\frac{i\,e^{i\,v}}{2}-\frac{i\,e^{-i\,v}}{1-e^{2\,i\,w}}-\frac{i\,e^{i\,v}}{1-e^{2\,i\,w}}$

Rule: If $b^2-d^2\neq 0$, then

$$\int \mathrm{Sin}[a+b\,x]\,\mathrm{Tan}[c+d\,x]\,dx \longrightarrow \int\left(\frac{e^{-i\,(a+b\,x)}}{2}-\frac{e^{i\,(a+b\,x)}}{2}-\frac{e^{-i\,(a+b\,x)}}{1+e^{2\,i\,(c+d\,x)}}+\frac{e^{i\,(a+b\,x)}}{1+e^{2\,i\,(c+d\,x)}}\right)dx$$

## Program code:

```
Int[Sin[a_.+b_.*x_]*Tan[c_.+d_.*x_],x_Symbol] :=
  Int[E^(-I*(a+b*x))/2 - E^(I*(a+b*x))/2 - E^(-I*(a+b*x))/(1+E^(2*I*(c+d*x))) + E^(I*(a+b*x))/(1+E^(2*I*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

```
Int[Cos[a_.+b_.*x_]*Cot[c_.+d_.*x_],x_Symbol] :=
  Int[I*E^(-I*(a+b*x))/2 + I*E^(I*(a+b*x))/2 - I*E^(-I*(a+b*x))/(1-E^(2*I*(c+d*x))) - I*E^(I*(a+b*x))/(1-E^(2*I*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

**4:** $\int \mathrm{Sin}[a+b\,x]\,\mathrm{Cot}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

## Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[v]\,\mathrm{Cot}[w] == -\frac{e^{-i\,v}}{2}+\frac{e^{i\,v}}{2}+\frac{e^{-i\,v}}{1-e^{2\,i\,w}}-\frac{e^{i\,v}}{1-e^{2\,i\,w}}$

Basis: $\mathrm{Cos}[v]\,\mathrm{Tan}[w] == -\frac{i\,e^{-i\,v}}{2}-\frac{i\,e^{i\,v}}{2}+\frac{i\,e^{-i\,v}}{1+e^{2\,i\,w}}+\frac{i\,e^{i\,v}}{1+e^{2\,i\,w}}$

Rule: If $b^2-d^2\neq 0$, then

$$\int \mathrm{Sin}[a+b\,x]\,\mathrm{Cot}[c+d\,x]\,dx \longrightarrow \int\left(-\frac{e^{-i\,(a+b\,x)}}{2}+\frac{e^{i\,(a+b\,x)}}{2}+\frac{e^{-i\,(a+b\,x)}}{1-e^{2\,i\,(c+d\,x)}}-\frac{e^{i\,(a+b\,x)}}{1-e^{2\,i\,(c+d\,x)}}\right)dx$$

## Program code:

```
Int[Sin[a_.+b_.*x_]*Cot[c_.+d_.*x_],x_Symbol] :=
  Int[-E^(-I*(a+b*x))/2 + E^(I*(a+b*x))/2 + E^(-I*(a+b*x))/(1-E^(2*I*(c+d*x))) - E^(I*(a+b*x))/(1-E^(2*I*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

```
Int[Cos[a_.+b_.*x_]*Tan[c_.+d_.*x_],x_Symbol] :=
  Int[-I*E^(-I*(a+b*x))/2 - I*E^(I*(a+b*x))/2 + I*E^(-I*(a+b*x))/(1+E^(2*I*(c+d*x))) + I*E^(I*(a+b*x))/(1+E^(2*I*(c+d*x))),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

1: $\int \text{Sin}\left[\frac{a}{c+d\,x}\right]^n dx$ when $n \in \mathbb{Z}^+$

## Derivation: Integration by substitution

## Basis: $F\left[\frac{a}{c+d\,x}\right] == -\frac{1}{d}\,\text{Subst}\left[\frac{F[a\,x]}{x^2},\,x,\,\frac{1}{c+d\,x}\right]\partial_x \frac{1}{c+d\,x}$

## Rule: If $n \in \mathbb{Z}^+$, then

$$\int \text{Sin}\left[\frac{a}{c+d\,x}\right]^n dx \longrightarrow -\frac{1}{d}\,\text{Subst}\left[\int \frac{\text{Sin}[a\,x]^n}{x^2}\,dx,\,x,\,\frac{1}{c+d\,x}\right]$$

## Program code:

```
Int[Sin[a_./(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Sin[a*x]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,c,d},x] && IGtQ[n,0]
```

```
Int[Cos[a_./(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Cos[a*x]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,c,d},x] && IGtQ[n,0]
```

2. $\int \text{Sin}\left[\frac{a+b\,x}{c+d\,x}\right]^n dx$ when $n \in \mathbb{Z}^+$

1: $\int \text{Sin}\left[\frac{a+b\,x}{c+d\,x}\right]^n dx$ when $n \in \mathbb{Z}^+ \land b\,c-a\,d \neq 0$

## Derivation: Integration by substitution

Basis: $F\left[\frac{a+b\,x}{c+d\,x}\right] == -\frac{1}{d}\,\text{Subst}\left[\frac{F\left[\frac{b}{d}-\frac{(b\,c-a\,d)\,x}{d}\right]}{x^2},\,x,\,\frac{1}{c+d\,x}\right]\partial_x\frac{1}{c+d\,x}$

Rule: If $n \in \mathbb{Z}^+ \land b\,c-a\,d \neq 0$, then

$$\int \text{Sin}\left[\frac{a+b\,x}{c+d\,x}\right]^n dx \to -\frac{1}{d}\,\text{Subst}\left[\int \frac{\text{Sin}\left[\frac{b}{d}-\frac{(b\,c-a\,d)\,x}{d}\right]^n}{x^2}\,dx,\,x,\,\frac{1}{c+d\,x}\right]$$

Program code:

```
Int[Sin[e_.*(a_.+b_.*x_)/(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Sin[b*e/d-e*(b*c-a*d)*x/d]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,0] && NeQ[b*c-a*d,0]
```

```
Int[Cos[e_.*(a_.+b_.*x_)/(c_.+d_.*x_)]^n_.,x_Symbol] :=
  -1/d*Subst[Int[Cos[b*e/d-e*(b*c-a*d)*x/d]^n/x^2,x],x,1/(c+d*x)] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,0] && NeQ[b*c-a*d,0]
```

**2:** $\int \text{Sin}[u]^n\,dx$ when $n \in \mathbb{Z}^+ \land u == \frac{a+b\,x}{c+d\,x}$

Derivation: Algebraic normalization

Rule: If $n \in \mathbb{Z}^+ \land u == \frac{a+b\,x}{c+d\,x}$, then

$$\int \text{Sin}[u]^n\,dx \to \int \text{Sin}\left[\frac{a+b\,x}{c+d\,x}\right]^n dx$$

Program code:

```
Int[Sin[u_]^n_.,x_Symbol] :=
  Module[{lst=QuotientOfLinearsParts[u,x]},
  Int[Sin[(lst[[1]]+lst[[2]]*x)/(lst[[3]]+lst[[4]]*x)]^n,x]] /;
IGtQ[n,0] && QuotientOfLinearsQ[u,x]
```

```
Int[Cos[u_]^n_.,x_Symbol] :=
  Module[{lst=QuotientOfLinearsParts[u,x]},
  Int[Cos[(lst[[1]]+lst[[2]]*x)/(lst[[3]]+lst[[4]]*x)]^n,x]] /;
IGtQ[n,0] && QuotientOfLinearsQ[u,x]
```

3. $\int u\,\mathrm{Sin}[v]^p\,\mathrm{Trig}[w]^q\,dx$

1. $\int u\,\mathrm{Sin}[v]^p\,\mathrm{Sin}[w]^q\,dx$

1: $\int u\,\mathrm{Sin}[v]^p\,\mathrm{Sin}[w]^q\,dx$ when $w == v$

## Derivation: Algebraic simplification

## Rule: If $w == v$, then

$$\int u\,\mathrm{Sin}[v]^p\,\mathrm{Sin}[w]^q\,dx \longrightarrow \int u\,\mathrm{Sin}[v]^{p+q}\,dx$$

## Program code:

```
Int[u_.*Sin[v_]^p_.*Sin[w_]^q_.,x_Symbol] :=
  Int[u*Sin[v]^(p+q),x] /;
EqQ[w,v]
```

```
Int[u_.*Cos[v_]^p_.*Cos[w_]^q_.,x_Symbol] :=
  Int[u*Cos[v]^(p+q),x] /;
EqQ[w,v]
```

**2:** $\int \text{Sin}[v]^p \, \text{Sin}[w]^q \, dx$ when $(p \mid q) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

## Rule: If $(p \mid q) \in \mathbb{Z}^+$, then

$$\int \text{Sin}[v]^p \, \text{Sin}[w]^q \, dx \longrightarrow \int \text{TrigReduce}\left[\text{Sin}[v]^p \, \text{Sin}[w]^q\right] dx$$

## Program code:

```
Int[Sin[v_]^p_.*Sin[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[Sin[v]^p*Sin[w]^q,x],x] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x]) && IGtQ[p,0] && IGtQ[q,0]
```

```
Int[Cos[v_]^p_.*Cos[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[Cos[v]^p*Cos[w]^q,x],x] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x]) && IGtQ[p,0] && IGtQ[q,0]
```

3: $\int x^m \operatorname{Sin}[v]^p \operatorname{Sin}[w]^q \, dx$ when $(m \mid p \mid q) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

## Rule: If $(m \mid p \mid q) \in \mathbb{Z}^+$, then

$$\int x^m \operatorname{Sin}[v]^p \operatorname{Sin}[w]^q \, dx \longrightarrow \int x^m \operatorname{TrigReduce}\left[\operatorname{Sin}[v]^p \operatorname{Sin}[w]^q\right] dx$$

## Program code:

```
Int[x_^m_.*Sin[v_]^p_.*Sin[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^m,Sin[v]^p*Sin[w]^q,x],x] /;
IGtQ[m,0] && IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

```
Int[x_^m_.*Cos[v_]^p_.*Cos[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^m,Cos[v]^p*Cos[w]^q,x],x] /;
IGtQ[m,0] && IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

2. $\int u\,\text{Sin}[v]^p\,\text{Cos}[w]^q\,dx$

1: $\int u\,\text{Sin}[v]^p\,\text{Cos}[w]^p\,dx$ when $w == v \land p \in \mathbb{Z}$

## Derivation: Algebraic simplification

## Basis: $\text{Sin}[z]\,\text{Cos}[z] == \frac{1}{2}\,\text{Sin}[2\,z]$

## Rule: If $w == v \land p \in \mathbb{Z}$, then

$$\int u\,\text{Sin}[v]^p\,\text{Cos}[w]^p\,dx \longrightarrow \frac{1}{2^p}\int u\,\text{Sin}[2\,v]^p\,dx$$

## Program code:

```
Int[u_.*Sin[v_]^p_.*Cos[w_]^p_.,x_Symbol] :=
  1/2^p*Int[u*Sin[2*v]^p,x] /;
EqQ[w,v] && IntegerQ[p]
```

**2:** $\int \mathrm{Sin}[v]^p \, \mathrm{Cos}[w]^q \, dx$ when $(p \mid q) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

## Rule: If $(p \mid q) \in \mathbb{Z}^+$, then

$$\int \mathrm{Sin}[v]^p \, \mathrm{Cos}[w]^q \, dx \longrightarrow \int \mathrm{TrigReduce}\left[\mathrm{Sin}[v]^p \, \mathrm{Cos}[w]^q\right] dx$$

## Program code:

```
Int[Sin[v_]^p_.*Cos[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[Sin[v]^p*Cos[w]^q,x],x] /;
IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

**3:** $\int x^m \, \mathrm{Sin}[v]^p \, \mathrm{Cos}[w]^q \, dx$ when $(m \mid p \mid q) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

## Rule: If $(m \mid p \mid q) \in \mathbb{Z}^+$, then

$$\int x^m \, \mathrm{Sin}[v]^p \, \mathrm{Cos}[w]^q \, dx \longrightarrow \int x^m \, \mathrm{TrigReduce}\left[\mathrm{Sin}[v]^p \, \mathrm{Cos}[w]^q\right] dx$$

## Program code:

```
Int[x_^m_.*Sin[v_]^p_.*Cos[w_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^m,Sin[v]^p*Cos[w]^q,x],x] /;
IGtQ[m,0] && IGtQ[p,0] && IGtQ[q,0] && (PolynomialQ[v,x] && PolynomialQ[w,x] || BinomialQ[{v,w},x] && IndependentQ[Cancel[v/w],x])
```

3. $\int u \, \mathrm{Sin}[v]^p \, \mathrm{Tan}[w]^q \, dx$

1: $\int \mathrm{Sin}[v] \, \mathrm{Tan}[w]^n \, dx$ when $n > 0 \land x \notin v - w \land w \neq v$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[v] \, \mathrm{Tan}[w] == -\mathrm{Cos}[v] + \mathrm{Cos}[v - w] \, \mathrm{Sec}[w]$

Basis: $\mathrm{Cos}[v] \, \mathrm{Cot}[w] == -\mathrm{Sin}[v] + \mathrm{Cos}[v - w] \, \mathrm{Csc}[w]$

Rule: If $n > 0 \land x \notin v - w \land w \neq v$, then

$$\int \mathrm{Sin}[v] \, \mathrm{Tan}[w]^n \, dx \longrightarrow -\int \mathrm{Cos}[v] \, \mathrm{Tan}[w]^{n-1} \, dx + \mathrm{Cos}[v - w] \int \mathrm{Sec}[w] \, \mathrm{Tan}[w]^{n-1} \, dx$$

Program code:

```
Int[Sin[v_]*Tan[w_]^n_.,x_Symbol] :=
  -Int[Cos[v]*Tan[w]^(n-1),x] + Cos[v-w]*Int[Sec[w]*Tan[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

```
Int[Cos[v_]*Cot[w_]^n_.,x_Symbol] :=
  -Int[Sin[v]*Cot[w]^(n-1),x] + Cos[v-w]*Int[Csc[w]*Cot[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

4. $\int u\,\text{Sin}[v]^p\,\text{Cot}[w]^q\,dx$

**1:** $\int \text{Sin}[v]\,\text{Cot}[w]^n\,dx$ when $n>0 \land x\notin v-w \land w\neq v$

Derivation: Algebraic expansion

Basis: $\text{Sin}[v]\,\text{Cot}[w] == \text{Cos}[v]+\text{Sin}[v-w]\,\text{Csc}[w]$

Basis: $\text{Cos}[v]\,\text{Tan}[w] == \text{Sin}[v]-\text{Sin}[v-w]\,\text{Sec}[w]$

Rule: If $n>0 \land x\notin v-w \land w\neq v$, then

$$\int \text{Sin}[v]\,\text{Cot}[w]^n\,dx \to \int \text{Cos}[v]\,\text{Cot}[w]^{n-1}\,dx+\text{Sin}[v-w]\int \text{Csc}[w]\,\text{Cot}[w]^{n-1}\,dx$$

Program code:

```
Int[Sin[v_]*Cot[w_]^n_.,x_Symbol] :=
  Int[Cos[v]*Cot[w]^(n-1),x] + Sin[v-w]*Int[Csc[w]*Cot[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

```
Int[Cos[v_]*Tan[w_]^n_.,x_Symbol] :=
  Int[Sin[v]*Tan[w]^(n-1),x] - Sin[v-w]*Int[Sec[w]*Tan[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

5. $\int u\,\mathrm{Sin}[v]^p\,\mathrm{Sec}[w]^q\,dx$

**1:** $\int \mathrm{Sin}[v]\,\mathrm{Sec}[w]^n\,dx$ when $n>0\land x\notin v-w\land w\neq v$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[v]\,\mathrm{Sec}[w]==\mathrm{Cos}[v-w]\,\mathrm{Tan}[w]+\mathrm{Sin}[v-w]$

Basis: $\mathrm{Cos}[v]*\mathrm{Csc}[w]==\mathrm{Cos}[v-w]*\mathrm{Cot}[w]-\mathrm{Sin}[v-w]$

Rule: If $n>0\land x\notin v-w\land w\neq v$, then

$$\int \mathrm{Sin}[v]\,\mathrm{Sec}[w]^n\,dx\longrightarrow \mathrm{Cos}[v-w]\int \mathrm{Tan}[w]\,\mathrm{Sec}[w]^{n-1}\,dx+\mathrm{Sin}[v-w]\int \mathrm{Sec}[w]^{n-1}\,dx$$

Program code:

```
Int[Sin[v_]*Sec[w_]^n_.,x_Symbol] :=
  Cos[v-w]*Int[Tan[w]*Sec[w]^(n-1),x] + Sin[v-w]*Int[Sec[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

```
Int[Cos[v_]*Csc[w_]^n_.,x_Symbol] :=
  Cos[v-w]*Int[Cot[w]*Csc[w]^(n-1),x] - Sin[v-w]*Int[Csc[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

6. $\int u \operatorname{Sin}[v]^p \operatorname{Csc}[w]^q \, dx$

**1:** $\int \operatorname{Sin}[v] \operatorname{Csc}[w]^n \, dx$ when $n > 0 \land x \notin v - w \land w \neq v$

Derivation: Algebraic expansion

Basis: $\operatorname{Sin}[v] \operatorname{Csc}[w] == \operatorname{Sin}[v-w] \operatorname{Cot}[w] + \operatorname{Cos}[v-w]$

Basis: $\operatorname{Cos}[v] \operatorname{Sec}[w] == -\operatorname{Sin}[v-w] \operatorname{Tan}[w] + \operatorname{Cos}[v-w]$

Rule: If $n > 0 \land x \notin v - w \land w \neq v$, then

$$\int \operatorname{Sin}[v] \operatorname{Csc}[w]^n \, dx \longrightarrow \operatorname{Sin}[v-w] \int \operatorname{Cot}[w] \operatorname{Csc}[w]^{n-1} \, dx + \operatorname{Cos}[v-w] \int \operatorname{Csc}[w]^{n-1} \, dx$$

Program code:

```
Int[Sin[v_]*Csc[w_]^n_.,x_Symbol] :=
  Sin[v-w]*Int[Cot[w]*Csc[w]^(n-1),x] + Cos[v-w]*Int[Csc[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

```
Int[Cos[v_]*Sec[w_]^n_.,x_Symbol] :=
  -Sin[v-w]*Int[Tan[w]*Sec[w]^(n-1),x] + Cos[v-w]*Int[Sec[w]^(n-1),x] /;
GtQ[n,0] && FreeQ[v-w,x] && NeQ[w,v]
```

**4:** $\int (e + f x)^m (a + b \operatorname{Sin}[c + d x] \operatorname{Cos}[c + d x])^n \, dx$

## Derivation: Algebraic simplification

## Basis: $\operatorname{Sin}[z] \operatorname{Cos}[z] == \frac{1}{2} \operatorname{Sin}[2 z]$

## Rule:

$$\int (e + f x)^m (a + b \operatorname{Sin}[c + d x] \operatorname{Cos}[c + d x])^n \, dx \longrightarrow \int (e + f x)^m \left(a + \frac{1}{2} b \operatorname{Sin}[2 c + 2 d x]\right)^n dx$$

## Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_+b_.*Sin[c_.+d_.*x_]*Cos[c_.+d_.*x_])^n_.,x_Symbol] :=
  Int[(e+f*x)^m*(a+b*Sin[2*c+2*d*x]/2)^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

**5:** $\int x^m \left(a+b\,\mathrm{Sin}[c+d\,x]^2\right)^n dx$ when $a+b\neq 0\land(m\mid n)\in\mathbb{Z}\land m>0\land n<0$

Derivation: Algebraic simplification

Basis: $\mathrm{Sin}[z]^2 == \frac{1}{2}\,(1-\mathrm{Cos}[2\,z])$

Note: This rule should be replaced with rules that directly reduce the integrand rather than transforming it using trig power expansion!

Rule: If $a+b\neq 0\land(m\mid n)\in\mathbb{Z}\land m>0\land n<0$, then

$$\int x^m \left(a+b\,\mathrm{Sin}[c+d\,x]^2\right)^n dx\ \longrightarrow\ \frac{1}{2^n}\int x^m\,(2\,a+b-b\,\mathrm{Cos}[2\,c+2\,d\,x])^n\,dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*Sin[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  1/2^n*Int[x^m*(2*a+b-b*Cos[2*c+2*d*x])^n,x] /;
FreeQ[{a,b,c,d},x] && NeQ[a+b,0] && IGtQ[m,0] && ILtQ[n,0] && (EqQ[n,-1] || EqQ[m,1] && EqQ[n,-2])
```

```
Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  1/2^n*Int[x^m*(2*a+b+b*Cos[2*c+2*d*x])^n,x] /;
FreeQ[{a,b,c,d},x] && NeQ[a+b,0] && IGtQ[m,0] && ILtQ[n,0] && (EqQ[n,-1] || EqQ[m,1] && EqQ[n,-2])
```

**6:** $\int \frac{(f+g\,x)^m}{a+b\,\mathrm{Cos}[d+e\,x]^2+c\,\mathrm{Sin}[d+e\,x]^2}\,dx$ when $m\in\mathbb{Z}^+\land a+b\neq 0\land a+c\neq 0$

Derivation: Algebraic simplification

Basis: $a+b\,\mathrm{Cos}[z]^2+c\,\mathrm{Sin}[z]^2 == \frac{1}{2}\,(2\,a+b+c+(b-c)\,\mathrm{Cos}[2\,z])$

Rule: If $m\in\mathbb{Z}^+\land a+b\neq 0\land a+c\neq 0$, then

$$\int \frac{(f+g\,x)^m}{a+b\,\text{Cos}[d+e\,x]^2+c\,\text{Sin}[d+e\,x]^2}\,dx \longrightarrow 2\int \frac{(f+g\,x)^m}{2\,a+b+c+(b-c)\,\text{Cos}[2\,d+2\,e\,x]}\,dx$$

## Program code:

```
Int[(f_.+g_.*x_)^m_./(a_.+b_.*Cos[d_.+e_.*x_]^2+c_.*Sin[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(2*a+b+c+(b-c)*Cos[2*d+2*e*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[m,0] && NeQ[a+b,0] && NeQ[a+c,0]
```

```
Int[(f_.+g_.*x_)^m_.*Sec[d_.+e_.*x_]^2/(b_+c_.*Tan[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(b+c+(b-c)*Cos[2*d+2*e*x]),x] /;
FreeQ[{b,c,d,e,f,g},x] && IGtQ[m,0]
```

```
Int[(f_.+g_.*x_)^m_.*Sec[d_.+e_.*x_]^2/(b_.+a_.*Sec[d_.+e_.*x_]^2+c_.*Tan[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(2*a+b+c+(b-c)*Cos[2*d+2*e*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[m,0] && NeQ[a+b,0] && NeQ[a+c,0]
```

```
Int[(f_.+g_.*x_)^m_.*Csc[d_.+e_.*x_]^2/(c_+b_.*Cot[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(b+c+(b-c)*Cos[2*d+2*e*x]),x] /;
FreeQ[{b,c,d,e,f,g},x] && IGtQ[m,0]
```

```
Int[(f_.+g_.*x_)^m_.*Csc[d_.+e_.*x_]^2/(c_.+b_.*Cot[d_.+e_.*x_]^2+a_.*Csc[d_.+e_.*x_]^2),x_Symbol] :=
  2*Int[(f+g*x)^m/(2*a+b+c+(b-c)*Cos[2*d+2*e*x]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[m,0] && NeQ[a+b,0] && NeQ[a+c,0]
```

**7:** $\int \frac{(e+f\,x)\,(A+B\,\text{Sin}[c+d\,x])}{(a+b\,\text{Sin}[c+d\,x])^2}\,dx$ when $a\,A-b\,B==0$

## Derivation: Integration by parts

Basis: If $a\,A-b\,B==0$, then $\frac{(A+B\,\text{Sin}[c+d\,x])}{(a+b\,\text{Sin}[c+d\,x])^2}==-\partial_x\frac{B\,\text{Cos}[c+d\,x]}{a\,d\,(a+b\,\text{Sin}[c+d\,x])}$

Rule: If $a\,A-b\,B==0$, then

$$\int \frac{(e+f\,x)\,(A+B\,\text{Sin}[c+d\,x])}{(a+b\,\text{Sin}[c+d\,x])^2}\,dx \longrightarrow -\frac{B\,(e+f\,x)\,\text{Cos}[c+d\,x]}{a\,d\,(a+b\,\text{Sin}[c+d\,x])}+\frac{B\,f}{a\,d}\int \frac{\text{Cos}[c+d\,x]}{a+b\,\text{Sin}[c+d\,x]}\,dx$$

## Program code:

```
Int[(e_.+f_.*x_)*(A_+B_.*Sin[c_.+d_.*x_])/(a_+b_.*Sin[c_.+d_.*x_])^2,x_Symbol] :=
  -B*(e+f*x)*Cos[c+d*x]/(a*d*(a+b*Sin[c+d*x])) +
  B*f/(a*d)*Int[Cos[c+d*x]/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && EqQ[a*A-b*B,0]
```

```
Int[(e_.+f_.*x_)*(A_+B_.*Cos[c_.+d_.*x_])/(a_+b_.*Cos[c_.+d_.*x_])^2,x_Symbol] :=
  B*(e+f*x)*Sin[c+d*x]/(a*d*(a+b*Cos[c+d*x])) -
  B*f/(a*d)*Int[Sin[c+d*x]/(a+b*Cos[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && EqQ[a*A-b*B,0]
```

8. $\int \frac{(b\,x)^m \operatorname{Sin}[a\,x]^n}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2}\,dx$ when $a\,c + d == 0 \land m == 2 - n$

**1:** $\int \frac{x^2}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2}\,dx$ when $a\,c + d == 0$

Derivation: Integration by parts

Basis: If $a\,c + d == 0$, then $\frac{x \operatorname{Sin}[a\,x]}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2} == \partial_x \frac{1}{a\,d\,(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])}$

Basis: If $a\,c + d == 0$, then $\partial_x \frac{x}{\operatorname{Sin}[a\,x]} == \frac{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])}{c \operatorname{Sin}[a\,x]^2}$

Rule: If $a\,c + d == 0$, then

$$\int \frac{x^2}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2}\,dx \to \frac{x}{a\,d \operatorname{Sin}[a\,x]\,(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])} + \frac{1}{d^2}\int \frac{1}{\operatorname{Sin}[a\,x]^2}\,dx$$

Program code:

```
Int[x_^2/(c_.*Sin[a_.*x_]+d_.*x_*Cos[a_.*x_])^2,x_Symbol] :=
  x/(a*d*Sin[a*x]*(c*Sin[a*x]+d*x*Cos[a*x])) + 1/d^2*Int[1/Sin[a*x]^2,x] /;
FreeQ[{a,c,d},x] && EqQ[a*c+d,0]
```

```
Int[x_^2/(c_.*Cos[a_.*x_]+d_.*x_*Sin[a_.*x_])^2,x_Symbol] :=
  -x/(a*d*Cos[a*x]*(c*Cos[a*x]+d*x*Sin[a*x])) + 1/d^2*Int[1/Cos[a*x]^2,x] /;
FreeQ[{a,c,d},x] && EqQ[a*c-d,0]
```

**2:** $\int \frac{\operatorname{Sin}[a\,x]^2}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2}\,dx$ when $a\,c + d == 0$

Derivation: Integration by parts

Basis: If $a\,c + d == 0$, then $\frac{b\,x \operatorname{Sin}[a\,x]}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2} == \partial_x \frac{b}{a\,d\,(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])}$

Basis: If $a c + d == 0 \wedge m == 2 - n$, then
$\partial_x \left((b x)^{m-1} \operatorname{Sin}[a x]^{n-1}\right) == -\frac{b (n-1)}{c} (b x)^{m-2} \operatorname{Sin}[a x]^{n-2} (c \operatorname{Sin}[a x] + d x \operatorname{Cos}[a x])$

Rule: If $a c + d == 0 \wedge m == 2 - n$, then

$$\int \frac{\operatorname{Sin}[a x]^2}{(c \operatorname{Sin}[a x] + d x \operatorname{Cos}[a x])^2} \, dx \longrightarrow \frac{1}{d^2 x} + \frac{\operatorname{Sin}[a x]}{a d x (d x \operatorname{Cos}[a x] + c \operatorname{Sin}[a x])}$$

Program code:

```
Int[Sin[a_.*x_]^2/(c_.*Sin[a_.*x_]+d_.*x_*Cos[a_.*x_])^2,x_Symbol] :=
  1/(d^2*x) + Sin[a*x]/(a*d*x*(d*x*Cos[a*x]+c*Sin[a*x])) /;
FreeQ[{a,c,d},x] && EqQ[a*c+d,0]
```

```
Int[Cos[a_.*x_]^2/(c_.*Cos[a_.*x_]+d_.*x_*Sin[a_.*x_])^2,x_Symbol] :=
  1/(d^2*x) - Cos[a*x]/(a*d*x*(d*x*Sin[a*x]+c*Cos[a*x])) /;
FreeQ[{a,c,d},x] && EqQ[a*c-d,0]
```

**3:** $\int \frac{(b\,x)^m \operatorname{Sin}[a\,x]^n}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2} \, dx$ when $a\,c + d == 0 \land m == 2 - n$

## Derivation: Integration by parts

Basis: If $a\,c + d == 0$, then $\frac{b\,x \operatorname{Sin}[a\,x]}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2} == \partial_x \frac{b}{a\,d\,(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])}$

Basis: If $a\,c + d == 0 \land m == 2 - n$, then
$\partial_x \left((b\,x)^{m-1} \operatorname{Sin}[a\,x]^{n-1}\right) == -\frac{b\,(n-1)}{c} (b\,x)^{m-2} \operatorname{Sin}[a\,x]^{n-2} (c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])$

Rule: If $a\,c + d == 0 \land m == 2 - n$, then

$$\int \frac{(b\,x)^m \operatorname{Sin}[a\,x]^n}{(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])^2} \, dx \rightarrow \frac{b\,(b\,x)^{m-1} \operatorname{Sin}[a\,x]^{n-1}}{a\,d\,(c \operatorname{Sin}[a\,x] + d\,x \operatorname{Cos}[a\,x])} - \frac{b^2\,(n-1)}{d^2} \int (b\,x)^{m-2} \operatorname{Sin}[a\,x]^{n-2} \, dx$$

## Program code:

```
Int[(b_.*x_)^m_*Sin[a_.*x_]^n_/(c_.*Sin[a_.*x_]+d_.*x_*Cos[a_.*x_])^2,x_Symbol] :=
  b*(b*x)^(m-1)*Sin[a*x]^(n-1)/(a*d*(c*Sin[a*x]+d*x*Cos[a*x])) -
  b^2*(n-1)/d^2*Int[(b*x)^(m-2)*Sin[a*x]^(n-2),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[a*c+d,0] && EqQ[m,2-n]
```

```
Int[(b_.*x_)^m_*Cos[a_.*x_]^n_/(c_.*Cos[a_.*x_]+d_.*x_*Sin[a_.*x_])^2,x_Symbol] :=
  -b*(b*x)^(m-1)*Cos[a*x]^(n-1)/(a*d*(c*Cos[a*x]+d*x*Sin[a*x])) -
  b^2*(n-1)/d^2*Int[(b*x)^(m-2)*Cos[a*x]^(n-2),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[a*c-d,0] && EqQ[m,2-n]
```

## Rule: If $a c + d == 0 \land m == n + 2$, then

$$\int \frac{(b x)^m \operatorname{Csc}[a x]^n}{(c \operatorname{Sin}[a x] + d x \operatorname{Cos}[a x])^2} \, dx \longrightarrow \frac{b (b x)^{m-1} \operatorname{Csc}[a x]^{n+1}}{a d (c \operatorname{Sin}[a x] + d x \operatorname{Cos}[a x])} + \frac{b^2 (n+1)}{d^2} \int (b x)^{m-2} \operatorname{Csc}[a x]^{n+2} \, dx$$

```
Int[(b_.*x_)^m_.*Csc[a_.*x_]^n_./(c_.*Sin[a_.*x_]+d_.*x_*Cos[a_.*x_])^2,x_Symbol] :=
  b*(b*x)^(m-1)*Csc[a*x]^(n+1)/(a*d*(c*Sin[a*x]+d*x*Cos[a*x])) +
  b^2*(n+1)/d^2*Int[(b*x)^(m-2)*Csc[a*x]^(n+2),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[a*c+d,0] && EqQ[m,n+2]
```

```
Int[(b_.*x_)^m_.*Sec[a_.*x_]^n_./(c_.*Cos[a_.*x_]+d_.*x_*Sin[a_.*x_])^2,x_Symbol] :=
  -b*(b*x)^(m-1)*Sec[a*x]^(n+1)/(a*d*(c*Cos[a*x]+d*x*Sin[a*x])) +
  b^2*(n+1)/d^2*Int[(b*x)^(m-2)*Sec[a*x]^(n+2),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[a*c-d,0] && EqQ[m,n+2]
```

9. $\int (g + h x)^p (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c + a d == 0 \land a^2 - b^2 == 0 \land (2 m \mid n - m) \in \mathbb{Z}$

1: $\int (g + h x)^p (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx$ when $b c + a d == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z} \land n - m \in \mathbb{Z}^+$

## Derivation: Algebraic simplification

## Basis: If $b c + a d == 0 \land a^2 - b^2 == 0$, then $(a + b \operatorname{Sin}[z]) (c + d \operatorname{Sin}[z]) == a c \operatorname{Cos}[z]^2$

## Rule: If $b c + a d == 0 \land a^2 - b^2 == 0 \land m \in \mathbb{Z} \land n - m \in \mathbb{Z}^+$, then

$$\int (g + h x)^p (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n \, dx \longrightarrow a^m c^m \int (g + h x)^p \operatorname{Cos}[e + f x]^{2 m} (c + d \operatorname{Sin}[e + f x])^{n-m} \, dx$$

## Program code:

```
Int[(g_.+h_.*x_)^p_.*(a_+b_.*Sin[e_.+f_.*x_])^m_.*(c_+d_.*Sin[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^m*c^m*Int[(g+h*x)^p*Cos[e+f*x]^(2*m)*(c+d*Sin[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] && IGtQ[n-m,0]
```

```
Int[(g_.+h_.*x_)^p_.*(a_+b_.*Cos[e_.+f_.*x_])^m_.*(c_+d_.*Cos[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^m*c^m*Int[(g+h*x)^p*Sin[e+f*x]^(2*m)*(c+d*Cos[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] && IGtQ[n-m,0]
```

2: $\int (g+h\,x)^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,p\in\mathbb{Z}\,\land\,2\,m\in\mathbb{Z}\,\land\,n-m\in\mathbb{Z}^+$

## Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $\partial_x\frac{(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^m}{\text{Cos}[e+f\,x]^{2\,m}}==0$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,p\in\mathbb{Z}\,\land\,2\,m\in\mathbb{Z}\,\land\,n-m\in\mathbb{Z}^+$, then

$$\int (g+h\,x)^p\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\left(\left(a^{\text{IntPart}[m]}\,c^{\text{IntPart}[m]}\,(a+b\,\text{Sin}[e+f\,x])^{\text{FracPart}[m]}\,(c+d\,\text{Sin}[e+f\,x])^{\text{FracPart}[m]}\right)/\text{Cos}[e+f\,x]^{2\,\text{FracPart}[m]}\right)$$
$$\int (g+h\,x)^p\,\text{Cos}[e+f\,x]^{2\,m}\,(c+d\,\text{Sin}[e+f\,x])^{n-m}\,dx$$

## Program code:

```
Int[(g_.+h_.*x_)^p_.*(a_+b_.*Sin[e_.+f_.*x_])^m_*(c_+d_.*Sin[e_.+f_.*x_])^n_,x_Symbol] :=
  a^IntPart[m]*c^IntPart[m]*(a+b*Sin[e+f*x])^FracPart[m]*(c+d*Sin[e+f*x])^FracPart[m]/Cos[e+f*x]^(2*FracPart[m])*
    Int[(g+h*x)^p*Cos[e+f*x]^(2*m)*(c+d*Sin[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[p] && IntegerQ[2*m] && IGeQ[n-m,0]
```

```
Int[(g_.+h_.*x_)^p_.*(a_+b_.*Cos[e_.+f_.*x_])^m_*(c_+d_.*Cos[e_.+f_.*x_])^n_,x_Symbol] :=
  a^IntPart[m]*c^IntPart[m]*(a+b*Cos[e+f*x])^FracPart[m]*(c+d*Cos[e+f*x])^FracPart[m]/Sin[e+f*x]^(2*FracPart[m])*
    Int[(g+h*x)^p*Sin[e+f*x]^(2*m)*(c+d*Cos[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[p] && IntegerQ[2*m] && IGeQ[n-m,0]
```

**10:** $\int \text{Sec}[v]^m\,(a+b\,\text{Tan}[v])^n\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\wedge m+n==0$

## Derivation: Algebraic simplification

## Basis: $\frac{a+b\,\text{Tan}[z]}{\text{Sec}[z]}==a\,\text{Cos}[z]+b\,\text{Sin}[z]$

## Rule: If $\frac{m-1}{2}\in\mathbb{Z}\wedge m+n==0$, then

$$\int \text{Sec}[v]^m\,(a+b\,\text{Tan}[v])^n\,dx\longrightarrow\int(a\,\text{Cos}[v]+b\,\text{Sin}[v])^n\,dx$$

## Program code:

```
Int[Sec[v_]^m_.*(a_+b_.*Tan[v_])^n_., x_Symbol] :=
  Int[(a*Cos[v]+b*Sin[v])^n,x] /;
FreeQ[{a,b},x] && IntegerQ[(m-1)/2] && EqQ[m+n,0]
```

```
Int[Csc[v_]^m_.*(a_+b_.*Cot[v_])^n_., x_Symbol] :=
  Int[(b*Cos[v]+a*Sin[v])^n,x] /;
FreeQ[{a,b},x] && IntegerQ[(m-1)/2] && EqQ[m+n,0]
```

**11:** $\int u \operatorname{Sin}[a+b x]^m \operatorname{Sin}[c+d x]^n \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

## Derivation: Algebraic expansion

## Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int u \operatorname{Sin}[a+b x]^m \operatorname{Sin}[c+d x]^n \, dx \to \int u \operatorname{TrigReduce}\left[\operatorname{Sin}[a+b x]^m \operatorname{Sin}[c+d x]^n\right] dx$$

## Program code:

```
Int[u_.*Sin[a_.+b_.*x_]^m_.*Sin[c_.+d_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[u,Sin[a+b*x]^m*Sin[c+d*x]^n,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0] && IGtQ[n,0]
```

```
Int[u_.*Cos[a_.+b_.*x_]^m_.*Cos[c_.+d_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[u,Cos[a+b*x]^m*Cos[c+d*x]^n,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0] && IGtQ[n,0]
```

**12:** $\int \text{Sec}[a+b\,x]\,\text{Sec}[c+d\,x]\,dx$ when $b^2-d^2==0\wedge b\,c-a\,d\neq 0$

## Derivation: Algebraic expansion

Basis: If $b^2-d^2==0\wedge b\,c-a\,d\neq 0$, then
$\text{Sec}[a+b\,x]\,\text{Sec}[c+d\,x]==-\text{Csc}\left[\frac{b\,c-a\,d}{d}\right]\text{Tan}[a+b\,x]+\text{Csc}\left[\frac{b\,c-a\,d}{b}\right]\text{Tan}[c+d\,x]$

Rule: If $b^2-d^2==0\wedge b\,c-a\,d\neq 0$, then

$$\int \text{Sec}[a+b\,x]\,\text{Sec}[c+d\,x]\,dx\to -\text{Csc}\left[\frac{b\,c-a\,d}{d}\right]\int \text{Tan}[a+b\,x]\,dx+\text{Csc}\left[\frac{b\,c-a\,d}{b}\right]\int \text{Tan}[c+d\,x]\,dx$$

## Program code:

```
Int[Sec[a_.+b_.*x_]*Sec[c_+d_.*x_],x_Symbol] :=
  -Csc[(b*c-a*d)/d]*Int[Tan[a+b*x],x] + Csc[(b*c-a*d)/b]*Int[Tan[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

```
Int[Csc[a_.+b_.*x_]*Csc[c_+d_.*x_],x_Symbol] :=
  Csc[(b*c-a*d)/b]*Int[Cot[a+b*x],x] - Csc[(b*c-a*d)/d]*Int[Cot[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

**13:** $\int \text{Tan}[a+b\,x]\,\text{Tan}[c+d\,x]\,dx$ when $b^2-d^2==0 \land b\,c-a\,d\neq 0$

## Derivation: Algebraic expansion

Basis: If $b^2-d^2==0$, then $\text{Tan}[a+b\,x]\,\text{Tan}[c+d\,x]==-\frac{b}{d}+\frac{b}{d}\,\text{Cos}\left[\frac{b\,c-a\,d}{d}\right]\,\text{Sec}[a+b\,x]\,\text{Sec}[c+d\,x]$

Rule: If $b^2-d^2==0 \land b\,c-a\,d\neq 0$, then

$$\int \text{Tan}[a+b\,x]\,\text{Tan}[c+d\,x]\,dx \longrightarrow -\frac{b\,x}{d}+\frac{b}{d}\,\text{Cos}\left[\frac{b\,c-a\,d}{d}\right]\int \text{Sec}[a+b\,x]\,\text{Sec}[c+d\,x]\,dx$$

## Program code:

```
Int[Tan[a_.+b_.*x_]*Tan[c_+d_.*x_],x_Symbol] :=
  -b*x/d + b/d*Cos[(b*c-a*d)/d]*Int[Sec[a+b*x]*Sec[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

```
Int[Cot[a_.+b_.*x_]*Cot[c_+d_.*x_],x_Symbol] :=
  -b*x/d + Cos[(b*c-a*d)/d]*Int[Csc[a+b*x]*Csc[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b^2-d^2,0] && NeQ[b*c-a*d,0]
```

**14:** $\int u\,(a\,\text{Cos}[v]+b\,\text{Sin}[v])^n\,dx$ when $a^2+b^2==0$

## Derivation: Algebraic simplification

Basis: If $a^2+b^2==0$, then $a\,\text{Cos}[z]+b\,\text{Sin}[z]==a\,e^{-\frac{a z}{b}}$

Rule: If $a^2+b^2==0$, then

$$\int u\,(a\,\text{Cos}[v]+b\,\text{Sin}[v])^n\,dx\;\longrightarrow\;\int u\left(a\,e^{-\frac{a v}{b}}\right)^n\,dx$$

## Program code:

```
Int[u_.*(a_.*Cos[v_]+b_.*Sin[v_])^n_.,x_Symbol] :=
  Int[u*(a*E^(-a/b*v))^n,x] /;
FreeQ[{a,b,n},x] && EqQ[a^2+b^2,0]
```

15. $\int u \operatorname{Sin}\left[d \left(a + b \operatorname{Log}\left[c\, x^n\right]\right)^2\right] dx$

1: $\int \operatorname{Sin}\left[d \left(a + b \operatorname{Log}\left[c\, x^n\right]\right)^2\right] dx$

Derivation: Algebraic expansion

Basis: $\operatorname{Sin}[z] = \frac{i}{2} e^{-i z} - \frac{i}{2} e^{i z}$

Rule:

$$\int \operatorname{Sin}\left[d \left(a + b \operatorname{Log}\left[c\, x^n\right]\right)^2\right] dx \to \frac{i}{2} \int e^{-i d \left(a+b \operatorname{Log}\left[c\, x^n\right]\right)^2} dx - \frac{i}{2} \int e^{i d \left(a+b \operatorname{Log}\left[c\, x^n\right]\right)^2} dx$$

Program code:

```
Int[Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  I/2*Int[E^(-I*d*(a+b*Log[c*x^n])^2),x] - I/2*Int[E^(I*d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  1/2*Int[E^(-I*d*(a+b*Log[c*x^n])^2),x] + 1/2*Int[E^(I*d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,n},x]
```

**2:** $\int (e\,x)^m \,\mathrm{Sin}\left[d\,\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2\right] dx$

## Derivation: Algebraic expansion

## Basis: $\mathrm{Sin}[z] = \frac{i}{2}\,e^{-i\,z} - \frac{i}{2}\,e^{i\,z}$

## Rule:

$$\int (e\,x)^m \,\mathrm{Sin}\left[d\,\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2\right] dx \,\longrightarrow\, \frac{i}{2}\int (e\,x)^m\, e^{-i\,d\,\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2} dx - \frac{i}{2}\int (e\,x)^m\, e^{i\,d\,\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)^2} dx$$

## Program code:

```
Int[(e_.*x_)^m_.*Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  I/2*Int[(e*x)^m*E^(-I*d*(a+b*Log[c*x^n])^2),x] - I/2*Int[(e*x)^m*E^(I*d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])^2],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(-I*d*(a+b*Log[c*x^n])^2),x] + 1/2*Int[(e*x)^m*E^(I*d*(a+b*Log[c*x^n])^2),x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```