

Rules for integrands of the form $(a \operatorname{Trg}[e + f x])^m (b \operatorname{Tan}[e + f x])^n$

1. $\int (a \operatorname{Sin}[e + f x])^m (b \operatorname{Tan}[e + f x])^n dx$

1: $\int (a \operatorname{Sin}[e + f x])^m (b \operatorname{Tan}[e + f x])^n dx$ when $m + n - 1 == 0$

Rule: If $m + n - 1 == 0$, then

$$\int (a \operatorname{Sin}[e + f x])^m (b \operatorname{Tan}[e + f x])^n dx \rightarrow - \frac{b (a \operatorname{Sin}[e + f x])^m (b \operatorname{Tan}[e + f x])^{n-1}}{f m}$$

Program code:

```
Int[(a_.*sin[e_+f_.*x_])^m_*(b_.*tan[e_+f_.*x_])^n_,x_Symbol]:=
  -b*(a*sin[e+f*x])^m*(b*tan[e+f*x])^(n-1)/(f*m) /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m+n-1,0]
```

2: $\int \sin[e+fx]^m \tan[e+fx]^n dx$ when $(m \mid n \mid \frac{m+n-1}{2}) \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $(m \mid n \mid \frac{m+n-1}{2}) \in \mathbb{Z}$, then

$$\sin[e+fx]^m \tan[e+fx]^n = -\frac{1}{f} \operatorname{Subst}\left[\frac{(1-x^2)^{\frac{m+n-1}{2}}}{x^n}, x, \cos[e+fx]\right] \partial_x \cos[e+fx]$$

Rule: If $(m \mid n \mid \frac{m+n-1}{2}) \in \mathbb{Z}$, then

$$\int \sin[e+fx]^m \tan[e+fx]^n dx \rightarrow -\frac{1}{f} \operatorname{Subst}\left[\int \frac{(1-x^2)^{\frac{m+n-1}{2}}}{x^n} dx, x, \cos[e+fx]\right]$$

Program code:

```
Int[sin[e_+f_*x_]^m_*tan[e_+f_*x_]^n_,x_Symbol] :=
  -1/f*Subst[Int[(1-x^2)^(m+n-1)/2/x^n,x],x,Cos[e+f*x]] /;
FreeQ[{e,f},x] && IntegersQ[m,n,(m+n-1)/2]
```

3: $\int \sin[e+fx]^m (b \tan[e+fx])^n dx$ when $\frac{m}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\sin[z]^2 = \frac{\tan[z]^2}{1+\tan[z]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\sin[e+fx]^m F[b \tan[e+fx]] = \frac{b}{f} \operatorname{Subst}\left[\frac{x^m F[x]}{(b^2+x^2)^{\frac{m}{2}+1}}, x, b \tan[e+fx]\right] \partial_x (b \tan[e+fx])$$

Rule: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\int \sin[e+fx]^m (b \tan[e+fx])^n dx \rightarrow \frac{b}{f} \operatorname{Subst} \left[\int \frac{x^{m+n}}{(b^2+x^2)^{\frac{m}{2}+1}} dx, x, b \tan[e+fx] \right]$$

Program code:

```
Int[sin[e_+f_*x_]^m_*(b_*tan[e_+f_*x_]^n_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
    b*ff/f*Subst[Int[(ff*x)^(m+n)/(b^2+ff^2*x^2)^(m/2+1),x],x,b*Tan[e+f*x]/ff] /;
    FreeQ[{b,e,f,n},x] && IntegerQ[m/2]
```

4: $\int (a \sin[e+fx])^m \tan[e+fx]^n dx$ when $\frac{n+1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n+1}{2} \in \mathbb{Z}$, then $\tan[e+fx]^n F[a \sin[e+fx]] = \frac{1}{f} \operatorname{Subst} \left[\frac{x^n F[x]}{(a^2-x^2)^{\frac{n+1}{2}}}, x, a \sin[e+fx] \right] \partial_x (a \sin[e+fx])$

Rule: If $\frac{n+1}{2} \in \mathbb{Z}$, then

$$\int (a \sin[e+fx])^m \tan[e+fx]^n dx \rightarrow \frac{1}{f} \operatorname{Subst} \left[\int \frac{x^{m+n}}{(a^2-x^2)^{\frac{n+1}{2}}} dx, x, a \sin[e+fx] \right]$$

Program code:

```
Int[(a_*sin[e_+f_*x_]^m_*tan[e_+f_*x_]^n_,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
    ff/f*Subst[Int[(ff*x)^(m+n)/(a^2-ff^2*x^2)^((n+1)/2),x],x,a*SIn[e+f*x]/ff] /;
    FreeQ[{a,e,f,m},x] && IntegerQ[(n+1)/2]
```

5. $\int (a \sin[e+fx])^m (b \tan[e+fx])^n dx$ when $n > 1$

1: $\int (a \sin[e+fx])^m (b \tan[e+fx])^n dx$ when $n > 1 \wedge m < -1$

Reference: G&R 2.510.6, CRC 334b

Reference: G&R 2.510.3, CRC 334a

Rule: If $n > 1 \wedge m < -1$, then

$$\int (a \sin[e+fx])^m (b \tan[e+fx])^n dx \rightarrow \frac{b (a \sin[e+fx])^{m+2} (b \tan[e+fx])^{n-1}}{a^2 f (n-1)} - \frac{b^2 (m+2)}{a^2 (n-1)} \int (a \sin[e+fx])^{m+2} (b \tan[e+fx])^{n-2} dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*sin[e+f*x])^(m+2)*(b*tan[e+f*x])^(n-1)/(a^2*f*(n-1)) -
  b^2*(m+2)/(a^2*(n-1))*Int[(a*sin[e+f*x])^(m+2)*(b*tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[n,1] && (LtQ[m,-1] || EqQ[m,-1] && EqQ[n,3/2]) && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n > 1$

Reference: G&R 2.510.1

Reference: G&R 2.510.4

Rule: If $n > 1$, then

$$\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{b (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n-1}}{f (n-1)} - \frac{b^2 (m+n-1)}{n-1} \int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n-2} dx$$

Program code:

```
Int[(a_.sin[e_.+f_.x_])^m_.*(b_.tan[e_.+f_.x_])^n_,x_Symbol] :=
  b*(a*sin[e+f*x])^m*(b*tan[e+f*x])^(n-1)/(f*(n-1)) -
  b^2*(m+n-1)/(n-1)*Int[(a*sin[e+f*x])^m*(b*tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f,m},x] && GtQ[n,1] && IntegerQ[2*m,2*n] && Not[GtQ[m,1] && Not[IntegerQ[(m-1)/2]]]
```

6. $\int (a \sin[e+fx])^m (b \tan[e+fx])^n dx$ when $n < -1$

1: $\int \frac{\sqrt{a \sin[e+fx]}}{(b \tan[e+fx])^{3/2}} dx$

Rule:

$$\int \frac{\sqrt{a \sin[e+fx]}}{(b \tan[e+fx])^{3/2}} dx \rightarrow \frac{2 \sqrt{a \sin[e+fx]}}{b f \sqrt{b \tan[e+fx]}} + \frac{a^2}{b^2} \int \frac{\sqrt{b \tan[e+fx]}}{(a \sin[e+fx])^{3/2}} dx$$

Program code:

```
Int[Sqrt[a_*sin[e_+f_*x_]]/(b_*tan[e_+f_*x_])^(3/2),x_Symbol]:=
  2*Sqrt[a*sin[e+f*x]]/(b*f*Sqrt[b*Tan[e+f*x]]) + a^2/b^2*Int[Sqrt[b*Tan[e+f*x]]/(a*sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,e,f},x]
```

$$\mathbf{2:} \int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \text{ when } n < -1 \wedge m > 1$$

Reference: G&R 2.510.5, CRC 323a

Reference: G&R 2.510.2, CRC 323b

Rule: If $n < -1 \wedge m > 1$, then

$$\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{(a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1}}{b f m} - \frac{a^2 (n+1)}{b^2 m} \int (a \operatorname{Sin}[e+fx])^{m-2} (b \operatorname{Tan}[e+fx])^{n+2} dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*m) -
  a^2*(n+1)/(b^2*m)*Int[(a*Sin[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && LtQ[n,-1] && GtQ[m,1] && IntegersQ[2*m,2*n]
```

$$\mathbf{3:} \int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \text{ when } n < -1 \wedge m+n+1 \neq 0$$

Reference: G&R 2.510.4

Reference: G&R 2.510.1

Rule: If $n < -1 \wedge m+n+1 \neq 0$, then

$$\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{(a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1}}{b f (m+n+1)} - \frac{n+1}{b^2 (m+n+1)} \int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+2} dx$$

Program code:

```
Int[(a_.sin[e_.+f_.x_])^m_.*(b_.tan[e_.+f_.x_])^n_,x_Symbol]:=
  (a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*(m+n+1)) -
  (n+1)/(b^2*(m+n+1))*Int[(a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f,m},x] && LtQ[n,-1] && NeQ[m+n+1,0] && IntegersQ[2*m,2*n] && Not[EqQ[n,-3/2] && EqQ[m,1]]
```


7: $\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $m > 1$

Reference: G&R 2.510.2, CRC 323b

Reference: G&R 2.510.5, CRC 323a

Rule: If $m > 1$, then

$$\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow -\frac{b (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n-1}}{f m} + \frac{a^2 (m+n-1)}{m} \int (a \operatorname{Sin}[e+fx])^{m-2} (b \operatorname{Tan}[e+fx])^n dx$$

Program code:

```
Int[(a_.*sin[e_+f_.*x_])^m_.*(b_.*tan[e_+f_.*x_])^n_.,x_Symbol]:=
-b*(a*Sin[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*m) +
a^2*(m+n-1)/m*Int[(a*Sin[e+f*x])^(m-2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && (GtQ[m,1] || EqQ[m,1] && EqQ[n,1/2]) && IntegersQ[2*m,2*n]
```

8: $\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $m < -1 \wedge m+n+1 \neq 0$

Reference: G&R 2.510.3, CRC 334a

Reference: G&R 2.510.6, CRC 334b

Rule: If $m < -1 \wedge m+n+1 \neq 0$, then

$$\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{b (a \operatorname{Sin}[e+fx])^{m+2} (b \operatorname{Tan}[e+fx])^{n-1}}{a^2 f (m+n+1)} + \frac{m+2}{a^2 (m+n+1)} \int (a \operatorname{Sin}[e+fx])^{m+2} (b \operatorname{Tan}[e+fx])^n dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  b*(a*Sin[e+f*x])^(m+2)*(b*Tan[e+f*x])^(n-1)/(a^2*f*(m+n+1)) +
  (m+2)/(a^2*(m+n+1))*Int[(a*Sin[e+f*x])^(m+2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && LtQ[m,-1] && NeQ[m+n+1,0] && IntegersQ[2*m,2*n]
```

9: $\int (a \operatorname{Sin}[e+fx])^m \operatorname{Tan}[e+fx]^n dx$ when $n \in \mathbb{Z} \wedge m \notin \mathbb{Z}$

Derivation: Algebraic normalization

Basis: $\operatorname{Tan}[z] == \frac{\operatorname{Sin}[z]}{\operatorname{Cos}[z]}$

Rule: If $n \in \mathbb{Z} \wedge m \notin \mathbb{Z}$, then

$$\int (a \operatorname{Sin}[e+fx])^m \operatorname{Tan}[e+fx]^n dx \rightarrow \frac{1}{a^n} \int \frac{(a \operatorname{Sin}[e+fx])^{m+n}}{\operatorname{Cos}[e+fx]^n} dx$$

Program code:

```
Int[(a_.*sin[e_+f_.*x_])^m_*tan[e_+f_.*x_]^n_,x_Symbol]:=
  1/a^n*Int[(a*Sin[e+f*x])^(m+n)/Cos[e+f*x]^n,x] /;
FreeQ[{a,e,f,m},x] && IntegerQ[n] && Not[IntegerQ[m]]
```

10. $\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n \notin \mathbb{Z}$

1: $\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n \notin \mathbb{Z} \wedge m < 0$

Derivation: Piecewise constant extraction

■ Basis: $\partial_x \frac{(\operatorname{Cos}[e+fx])^n (b \operatorname{Tan}[e+fx])^n}{(a \operatorname{Sin}[e+fx])^n} == 0$

Rule: If $n \notin \mathbb{Z} \wedge m < 0$, then

$$\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{(\operatorname{Cos}[e+fx])^n (b \operatorname{Tan}[e+fx])^n}{(a \operatorname{Sin}[e+fx])^n} \int \frac{(a \operatorname{Sin}[e+fx])^{m+n}}{\operatorname{Cos}[e+fx]^n} dx$$

Program code:

```
Int[(a_.*sin[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol]:=
  Cos[e+f*x]^n*(b*Tan[e+f*x])^n/(a*Sin[e+f*x])^n*Int[(a*Sin[e+f*x])^(m+n)/Cos[e+f*x]^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[n]] && (IltQ[m,0] || EqQ[m,1] && EqQ[n,-1/2] || IntegersQ[m-1/2,n-1/2])
```

2: $\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

— Basis: $\partial_x \frac{(\operatorname{Cos}[e+fx])^n (b \operatorname{Tan}[e+fx])^n}{(a \operatorname{Sin}[e+fx])^n} == 0$

Rule: If $n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Sin}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{a (\operatorname{Cos}[e+fx])^{n+1} (b \operatorname{Tan}[e+fx])^{n+1}}{b (a \operatorname{Sin}[e+fx])^{n+1}} \int \frac{(a \operatorname{Sin}[e+fx])^{m+n}}{\operatorname{Cos}[e+fx]^n} dx$$

Program code:

```
Int[(a_.*sin[e_+f_.*x_])^m_.*(b_.*tan[e_+f_.*x_])^n_,x_Symbol]:=
  a*Cos[e+f*x]^(n+1)*(b*Tan[e+f*x])^(n+1)/(b*(a*Sin[e+f*x])^(n+1))*Int[(a*Sin[e+f*x])^(m+n)/Cos[e+f*x]^n,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[n]]
```

2: $\int (a \operatorname{Cos}[e + f x])^m (b \operatorname{Tan}[e + f x])^n dx$ when $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

$$- \text{Basis: } \partial_x \left((a \operatorname{Cos}[e + f x])^m \left(\frac{\operatorname{Sec}[e+fx]}{a} \right)^m \right) == 0$$

Rule: If $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Cos}[e + f x])^m (b \operatorname{Tan}[e + f x])^n dx \rightarrow (a \operatorname{Cos}[e + f x])^{\operatorname{FracPart}[m]} \left(\frac{\operatorname{Sec}[e + f x]}{a} \right)^{\operatorname{FracPart}[m]} \int \frac{(b \operatorname{Tan}[e + f x])^n}{\left(\frac{\operatorname{Sec}[e+fx]}{a} \right)^m} dx$$

Program code:

```
Int[(a_.*cos[e_+f_.*x_])^m_*(b_.*tan[e_+f_.*x_])^n_,x_Symbol] :=
  (a*cos[e+f*x])^FracPart[m]*(Sec[e+f*x]/a)^FracPart[m]*Int[(b*Tan[e+f*x])^n/(Sec[e+f*x]/a)^m,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

3: $\int (a \operatorname{Cot}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x ((a \operatorname{Cot}[e+fx])^m (b \operatorname{Tan}[e+fx])^n) = 0$

Rule: If $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Cot}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow (a \operatorname{Cot}[e+fx])^m (b \operatorname{Tan}[e+fx])^m \int (b \operatorname{Tan}[e+fx])^{n-m} dx$$

Program code:

```
Int[(a_.*cot[e_+f_.*x_])^m_*(b_.*tan[e_+f_.*x_])^n_,x_Symbol] :=
  (a*Cot[e+f*x])^m*(b*Tan[e+f*x])^m*Int[(b*Tan[e+f*x])^(n-m),x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```

4. $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$

1: $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $m+n+1 = 0$

Rule: If $m+n+1 = 0$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow -\frac{(a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1}}{b f m}$$

Program code:

```
Int[(a_.*sec[e_+f_.*x_])^m_*(b_.*tan[e_+f_.*x_])^n_,x_Symbol] :=
  -(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*m) /;
FreeQ[{a,b,e,f,m,n},x] && EqQ[m+n+1,0]
```

2: $\int (a \operatorname{Sec}[e+fx])^m \operatorname{Tan}[e+fx]^n dx$ when $\frac{n-1}{2} \in \mathbb{Z} \wedge \neg \left(\frac{m}{2} \in \mathbb{Z} \wedge 0 < m < n+1 \right)$

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$\operatorname{Tan}[e+fx]^n \operatorname{F}[\operatorname{Sec}[e+fx]] = \frac{1}{f} \operatorname{Subst} \left[\frac{\operatorname{F}[x] (-1+x^2)^{\frac{n-1}{2}}}{x}, x, \operatorname{Sec}[e+fx] \right] \partial_x \operatorname{Sec}[e+fx]$$

Rule: If $\frac{n-1}{2} \in \mathbb{Z} \wedge \neg \left(\frac{m}{2} \in \mathbb{Z} \wedge 0 < m < n+1 \right)$, then

$$\int (a \operatorname{Sec}[e+fx])^m \operatorname{Tan}[e+fx]^n dx \rightarrow \frac{a}{f} \operatorname{Subst} \left[\int (ax)^{m-1} (-1+x^2)^{\frac{n-1}{2}} dx, x, \operatorname{Sec}[e+fx] \right]$$

Program code:

```
Int[(a_.*sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  a/f*Subst[Int[(a*x)^(m-1)*(-1+x^2)^((n-1)/2),x],x,Sec[e+f*x]] /;
FreeQ[{a,e,f,m},x] && IntegerQ[(n-1)/2] && Not[IntegerQ[m/2] && LtQ[0,m,n+1]]
```


3: $\int \operatorname{Sec}[e+fx]^m (b \operatorname{Tan}[e+fx])^n dx$ when $\frac{m}{2} \in \mathbb{Z} \wedge \neg \left(\frac{n-1}{2} \in \mathbb{Z} \wedge 0 < n < m-1 \right)$

Derivation: Integration by substitution

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\operatorname{Sec}[e+fx]^m F[\operatorname{Tan}[e+fx]] = \frac{1}{f} \operatorname{Subst}\left[F[x] (1+x^2)^{\frac{m}{2}-1}, x, \operatorname{Tan}[e+fx]\right] \partial_x \operatorname{Tan}[e+fx]$$

Rule: If $\frac{m}{2} \in \mathbb{Z} \wedge \neg \left(\frac{n-1}{2} \in \mathbb{Z} \wedge 0 < n < m-1 \right)$, then

$$\int \operatorname{Sec}[e+fx]^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{1}{f} \operatorname{Subst}\left[\int (bx)^n (1+x^2)^{\frac{m}{2}-1} dx, x, \operatorname{Tan}[e+fx]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_*(b_.*tan[e_.+f_.*x_] )^n_.,x_Symbol] :=
  1/f*Subst[Int[(b*x)^n*(1+x^2)^(m/2-1),x],x,Tan[e+f*x]] /;
FreeQ[{b,e,f,n},x] && IntegerQ[m/2] && Not[IntegerQ[(n-1)/2] && LtQ[0,n,m-1]]
```

4. $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n < -1$

1: $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n < -1 \wedge (m > 1 \vee m == 1 \wedge n == -\frac{3}{2})$

Reference: G&R 2.510.5, CRC 323a

Reference: G&R 2.510.2, CRC 323b

Rule: If $n < -1 \wedge (m > 1 \vee m == 1 \wedge n == -\frac{3}{2})$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{a^2 (a \operatorname{Sec}[e+fx])^{m-2} (b \operatorname{Tan}[e+fx])^{n+1}}{b f (n+1)} - \frac{a^2 (m-2)}{b^2 (n+1)} \int (a \operatorname{Sec}[e+fx])^{m-2} (b \operatorname{Tan}[e+fx])^{n+2} dx$$

Program code:

```
Int[(a_.*Sec[e_.+f_.*x_])^m_.*(b_.*Tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+1)/(b*f*(n+1)) -
  a^2*(m-2)/(b^2*(n+1))*Int[(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f},x] && LtQ[n,-1] && (GtQ[m,1] || EqQ[m,1] && EqQ[n,-3/2]) && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n < -1$

Reference: G&R 2.510.4

Reference: G&R 2.510.1

Rule: If $n < -1$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{(a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1}}{b f (n+1)} - \frac{m+n+1}{b^2 (n+1)} \int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+2} dx$$

Program code:

```
Int[(a_.*Sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*(n+1)) -
  (m+n+1)/(b^2*(n+1))*Int[(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+2),x] /;
FreeQ[{a,b,e,f,m},x] && LtQ[n,-1] && IntegersQ[2*m,2*n]
```

5. $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n > 1$

1: $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n > 1 \wedge (m < -1 \vee m == -1 \wedge n == \frac{3}{2})$

Reference: G&R 2.510.6, CRC 334b

Reference: G&R 2.510.3, CRC 334a

Rule: If $n > 1 \wedge (m < -1 \vee m == -1 \wedge n == \frac{3}{2})$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{b (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n-1}}{f m} - \frac{b^2 (n-1)}{a^2 m} \int (a \operatorname{Sec}[e+fx])^{m+2} (b \operatorname{Tan}[e+fx])^{n-2} dx$$

Program code:

```
Int[(a_.*Sec[e_.+f_.*x_])^m_*(b_.*Tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*m) -
  b^2*(n-1)/(a^2*m)*Int[(a*Sec[e+f*x])^(m+2)*(b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f},x] && GtQ[n,1] && (LtQ[m,-1] || EqQ[m,-1] && EqQ[n,3/2]) && IntegersQ[2*m,2*n]
```

2: $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $n > 1 \wedge m+n-1 \neq 0$

Reference: G&R 2.510.1

Reference: G&R 2.510.4

Rule: If $n > 1 \wedge m+n-1 \neq 0$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{b (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n-1}}{f (m+n-1)} - \frac{b^2 (n-1)}{m+n-1} \int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n-2} dx$$

Program code:

```
Int[(a_.*Sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n-1)/(f*(m+n-1)) -
  b^2*(n-1)/(m+n-1)*Int[(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n-2),x] /;
FreeQ[{a,b,e,f,m},x] && GtQ[n,1] && NeQ[m+n-1,0] && IntegersQ[2*m,2*n]
```

6: $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $m < -1$

Reference: G&R 2.510.3, CRC 334a

Reference: G&R 2.510.6, CRC 334b

Rule: If $m < -1$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow -\frac{(a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1}}{b f m} + \frac{m+n+1}{a^2 m} \int (a \operatorname{Sec}[e+fx])^{m+2} (b \operatorname{Tan}[e+fx])^n dx$$

Program code:

```
Int[(a_.*Sec[e_+f_.*x_])^m_*(b_.*tan[e_+f_.*x_])^n_,x_Symbol] :=
- (a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)/(b*f*m) +
(m+n+1)/(a^2*m)*Int[(a*Sec[e+f*x])^(m+2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && (LtQ[m,-1] || EqQ[m,-1] && EqQ[n,-1/2]) && IntegersQ[2*m,2*n]
```

7: $\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $m > 1 \wedge m+n-1 \neq 0$

Reference: G&R 2.510.2, CRC 323b

Reference: G&R 2.510.5, CRC 323a

Rule: If $m > 1 \wedge m+n-1 \neq 0$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow \frac{a^2 (a \operatorname{Sec}[e+fx])^{m-2} (b \operatorname{Tan}[e+fx])^{n+1}}{b f (m+n-1)} + \frac{a^2 (m-2)}{(m+n-1)} \int (a \operatorname{Sec}[e+fx])^{m-2} (b \operatorname{Tan}[e+fx])^n dx$$

Program code:

```
Int[(a_.*Sec[e_.+f_.*x_])^m_.*(b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a^2*(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^(n+1)/(b*f*(m+n-1)) +
  a^2*(m-2)/(m+n-1)*Int[(a*Sec[e+f*x])^(m-2)*(b*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,n},x] && (GtQ[m,1] || EqQ[m,1] && EqQ[n,1/2]) && NeQ[m+n-1,0] && IntegersQ[2*m,2*n]
```

8: $\int \frac{\operatorname{Sec}[e+fx]}{\sqrt{b \operatorname{Tan}[e+fx]}} dx$

Derivation: Piecewise constant extraction

■ Basis: $\partial_x \frac{\sqrt{\sin[e+fx]}}{\sqrt{\cos[e+fx]} \sqrt{b \operatorname{Tan}[e+fx]}} == 0$

Rule:

$$\int \frac{\operatorname{Sec}[e+fx]}{\sqrt{b \tan[e+fx]}} dx \rightarrow \frac{\sqrt{\sin[e+fx]}}{\sqrt{\cos[e+fx]} \sqrt{b \tan[e+fx]}} \int \frac{1}{\sqrt{\cos[e+fx]} \sqrt{\sin[e+fx]}} dx$$

Program code:

```
Int[sec[e_+f_*x_]/Sqrt[b_*tan[e_+f_*x_]],x_Symbol]:=
  Sqrt[Sin[e+f*x]]/(Sqrt[Cos[e+f*x]]*Sqrt[b*Tan[e+f*x]])*Int[1/(Sqrt[Cos[e+f*x]]*Sqrt[Sin[e+f*x]]),x] /;
FreeQ[{b,e,f},x]
```

9: $\int \frac{\sqrt{b \tan[e+fx]}}{\operatorname{Sec}[e+fx]} dx$

Derivation: Piecewise constant extraction

■ Basis: $\partial_x \frac{\sqrt{\cos[e+fx]} \sqrt{b \tan[e+fx]}}{\sqrt{\sin[e+fx]}} == 0$

Rule:

$$\int \frac{\sqrt{b \tan[e+fx]}}{\operatorname{Sec}[e+fx]} dx \rightarrow \frac{\sqrt{\cos[e+fx]} \sqrt{b \tan[e+fx]}}{\sqrt{\sin[e+fx]}} \int \frac{1}{\sqrt{\cos[e+fx]} \sqrt{\sin[e+fx]}} dx$$

Program code:

```
Int[Sqrt[b_*tan[e_+f_*x_]]/sec[e_+f_*x_],x_Symbol]:=
  Sqrt[Cos[e+f*x]]*Sqrt[b*Tan[e+f*x]]/Sqrt[Sin[e+f*x]]*Int[Sqrt[Cos[e+f*x]]*Sqrt[Sin[e+f*x]]],x] /;
FreeQ[{b,e,f},x]
```


10: $\int (a \operatorname{Sec}[e+fx])^m (b \tan[e+fx])^n dx$ when $n + \frac{1}{2} \in \mathbb{Z} \wedge m + \frac{1}{2} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

■ Basis: $\partial_x \frac{(b \tan[e+fx])^n}{(a \operatorname{Sec}[e+fx])^n (b \sin[e+fx])^n} = 0$

Rule: If $n + \frac{1}{2} \in \mathbb{Z} \wedge m + \frac{1}{2} \in \mathbb{Z}$, then

$$\int (a \operatorname{Sec}[e+fx])^m (b \tan[e+fx])^n dx \rightarrow \frac{a^{m+n} (b \tan[e+fx])^n}{(a \operatorname{Sec}[e+fx])^n (b \sin[e+fx])^n} \int \frac{(b \sin[e+fx])^n}{\cos[e+fx]^{m+n}} dx$$

Program code:

```
Int[(a_.*Sec[e_.+f_.*x_])^m*(b_.*tan[e_.+f_.*x_])^n,x_Symbol]:=
  a^(m+n)*(b*Tan[e+f*x])^n/((a*Sec[e+f*x])^n*(b*SIn[e+f*x])^n)*Int[(b*SIn[e+f*x])^n/Cos[e+f*x]^(m+n),x] /;
FreeQ[{a,b,e,f,m,n},x] && IntegerQ[n+1/2] && IntegerQ[m+1/2]
```

11: $\int (a \operatorname{Sec}[e+fx])^m (b \tan[e+fx])^n dx$ when $\frac{n-1}{2} \notin \mathbb{Z} \wedge \frac{m}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

■ Basis: $\partial_x \frac{(a \operatorname{Sec}[e+fx])^m (b \tan[e+fx])^{n+1} (\cos[e+fx]^2)^{\frac{m+n+1}{2}}}{(b \sin[e+fx])^{n+1}} = 0$

Basis: $\cos[e+fx] F[\sin[e+fx]] = \frac{1}{b f} \operatorname{Subst}\left[F\left[\frac{x}{b}\right], x, b \sin[e+fx]\right] \partial_x (b \sin[e+fx])$

Note: If $\frac{n}{2} \in \mathbb{Z}$, then $\frac{(a \operatorname{Sec}[e+fx])^m (b \tan[e+fx])^{n+1} (\cos[e+fx]^2)^{\frac{m+n+1}{2}}}{(b \sin[e+fx])^{n+1}} = (a \operatorname{Sec}[e+fx])^{m+1} (\cos[e+fx]^2)^{\frac{m+1}{2}}$

Note: If $\frac{n}{2} \in \mathbb{Z}$ and m is a third-integer integration of $\frac{x^n}{\left(1 - \frac{x^2}{b^2}\right)^{\frac{m+n+1}{2}}}$ results in a complicated antiderivative involving elliptic integrals and the imaginary unit.

Rule: If $\frac{n-1}{2} \notin \mathbb{Z} \wedge \frac{m}{2} \notin \mathbb{Z}$, then

$$\begin{aligned} \int (a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx &\rightarrow \frac{(a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1} (\operatorname{Cos}[e+fx]^2)^{\frac{m+n+1}{2}}}{(b \operatorname{Sin}[e+fx])^{n+1}} \int \frac{\operatorname{Cos}[e+fx] (b \operatorname{Sin}[e+fx])^n}{(1 - \operatorname{Sin}[e+fx]^2)^{\frac{m+n+1}{2}}} dx \\ &\rightarrow \frac{(a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1} (\operatorname{Cos}[e+fx]^2)^{\frac{m+n+1}{2}}}{b f (b \operatorname{Sin}[e+fx])^{n+1}} \operatorname{Subst}\left[\int \frac{x^n}{\left(1 - \frac{x^2}{b^2}\right)^{\frac{m+n+1}{2}}} dx, x, b \operatorname{Sin}[e+fx]\right] \\ &\rightarrow \frac{(a \operatorname{Sec}[e+fx])^m (b \operatorname{Tan}[e+fx])^{n+1} (\operatorname{Cos}[e+fx]^2)^{\frac{m+n+1}{2}}}{b f (n+1)} \operatorname{Hypergeometric2F1}\left[\frac{n+1}{2}, \frac{m+n+1}{2}, \frac{n+3}{2}, \operatorname{Sin}[e+fx]^2\right] \end{aligned}$$

Program code:

```
(* Int[(a_*sec[e_+f_*x_])^m_*(b_*tan[e_+f_*x_])^n_,x_Symbol]:=
(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)*(Cos[e+f*x]^2)^(m+n+1)/2/(b*f*(b*Sin[e+f*x])^(n+1))*
Subst[Int[x^n/(1-x^2/b^2)^(m+n+1)/2,x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[(n-1)/2]] && Not[IntegerQ[m/2]] *)
```

```
Int[(a_*sec[e_+f_*x_])^m_*(b_*tan[e_+f_*x_])^n_,x_Symbol]:=
(a*Sec[e+f*x])^m*(b*Tan[e+f*x])^(n+1)*(Cos[e+f*x]^2)^(m+n+1)/2/(b*f*(n+1))*
Hypergeometric2F1[(n+1)/2,(m+n+1)/2,(n+3)/2,Sin[e+f*x]^2] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[(n-1)/2]] && Not[IntegerQ[m/2]]
```

5: $\int (a \operatorname{Csc}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx$ when $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x ((a \operatorname{Csc}[e+fx])^m (a \operatorname{Sin}[e+fx])^m) = 0$

Rule: If $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z}$, then

$$\int (a \operatorname{Csc}[e+fx])^m (b \operatorname{Tan}[e+fx])^n dx \rightarrow (a \operatorname{Csc}[e+fx])^{\operatorname{FracPart}[m]} \left(\frac{\operatorname{Sin}[e+fx]}{a} \right)^{\operatorname{FracPart}[m]} \int \frac{(b \operatorname{Tan}[e+fx])^n}{\left(\frac{\operatorname{Sin}[e+fx]}{a} \right)^m} dx$$

Program code:

```
Int[(a_.*csc[e_+f_.*x_])^m_*(b_.*tan[e_+f_.*x_])^n_,x_Symbol] :=
  (a*Csc[e+f*x])^FracPart[m]*(Sin[e+f*x]/a)^FracPart[m]*Int[(b*Tan[e+f*x])^n/(Sin[e+f*x]/a)^m,x] /;
FreeQ[{a,b,e,f,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]]
```