

Rules for integrands of the form $u (a + b \operatorname{Log}[c (d + e x)^n])^p$

1: $\int (a + b \operatorname{Log}[c (d + e x)^n])^p dx$

Derivation: Integration by substitution

Rule:

$$\int (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow \frac{1}{e} \operatorname{Subst}\left[\int (a + b \operatorname{Log}[c x^n])^p dx, x, d + e x\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.])^p_. ,x_Symbol] :=  
  1/e*Subst[Int[(a+b*Log[c*x^n])^p,x],x,d+e*x] /;  
FreeQ[{a,b,c,d,e,n,p},x]
```

$$2. \int (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx$$

$$1. \int (f + g x)^q (a + b \log[c (d + e x)^n])^p dx$$

$$\text{1: } \int (f + g x)^q (a + b \log[c (d + e x)^n])^p dx \text{ when } e f - d g = 0$$

Derivation: Integration by substitution

Basis: If $e f - d g = 0$, then $(f + g x)^q F[d + e x] = \frac{1}{e} \text{Subst}\left[\left(\frac{f x}{d}\right)^q F[x], x, d + e x\right] \partial_x (d + e x)$

Rule: If $e f - d g = 0$, then

$$\int (f + g x)^q (a + b \log[c (d + e x)^n])^p dx \rightarrow \frac{1}{e} \text{Subst}\left[\int \left(\frac{f x}{d}\right)^q (a + b \log[c x^n])^p dx, x, d + e x\right]$$

Program code:

```
Int[(f_+g_.x_)^q_.*(a_+b_.*Log[c_.*(d_+e_.x_)^n_.])^p_.,x_Symbol] :=
  1/e*Subst[Int[(f*x/d)^q*(a+b*Log[c*x^n])^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q},x] && EqQ[e*f-d*g,0]
```

$$2. \int (f + g x)^q (a + b \log[c (d + e x)^n])^p dx \text{ when } e f - d g \neq 0$$

$$1. \int (f + g x)^q (a + b \log[c (d + e x)^n])^p dx \text{ when } e f - d g \neq 0 \wedge p > 0$$

$$1. \int (f + g x)^q (a + b \log[c (d + e x)^n]) dx \text{ when } e f - d g \neq 0$$

$$1. \int \frac{(a + b \log[c (d + e x)^n])}{f + g x} dx \text{ when } e f - d g \neq 0 \wedge p \in \mathbb{Z}^+$$

$$1. \int \frac{a + b \log[c (d + e x)]}{x} dx \text{ when } c d > 0$$

$$1: \int \frac{\text{Log}[c (d + e x^n)]}{x} dx \text{ when } c d = 1$$

Rule: If $c d = 1$, then

$$\int \frac{\text{Log}[c (d + e x^n)]}{x} dx \rightarrow -\frac{\text{PolyLog}[2, -c e x^n]}{n}$$

Program code:

```
Int[Log[c_.*(d_+e_.*x_^n_.)]/x_,x_Symbol] :=
  -PolyLog[2,-c*e*x^n]/n /;
FreeQ[{c,d,e,n},x] && EqQ[c*d,1]
```

$$2: \int \frac{a + b \text{Log}[c (d + e x)]}{x} dx \text{ when } c d > 0$$

Derivation: Algebraic expansion

Basis: If $c d > 0$, then $\text{Log}[c (d + e x)] = \text{Log}[c d] + \text{Log}\left[1 + \frac{e x}{d}\right]$

Rule: If $c d > 0$, then

$$\int \frac{a + b \text{Log}[c (d + e x)]}{x} dx \rightarrow (a + b \text{Log}[c d]) \text{Log}[x] + b \int \frac{\text{Log}\left[1 + \frac{e x}{d}\right]}{x} dx$$

Program code:

```
Int[(a_+b_.*Log[c_.*(d_+e_.*x_)])/x_,x_Symbol] :=
  (a+b*Log[c*d])*Log[x] + b*Int[Log[1+e*x/d]/x,x] /;
FreeQ[{a,b,c,d,e},x] && GtQ[c*d,0]
```

$$\mathbf{2:} \int \frac{a + b \operatorname{Log}[c (d + e x)]}{f + g x} dx \text{ when } e f - d g \neq 0 \wedge g + c (e f - d g) = 0$$

Derivation: Integration by substitution

Basis: If $g + c (e f - d g) = 0$, then $F[c (d + e x)] = \frac{1}{g} \operatorname{Subst}\left[F\left[1 + \frac{c e x}{g}\right], x, f + g x\right] \partial_x (f + g x)$

Rule: If $e f - d g \neq 0 \wedge g + c (e f - d g) = 0$, then

$$\int \frac{a + b \operatorname{Log}[c (d + e x)]}{f + g x} dx \rightarrow \frac{1}{g} \operatorname{Subst}\left[\int \frac{a + b \operatorname{Log}\left[1 + \frac{c e x}{g}\right]}{x} dx, x, f + g x\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.x_)])/(f_.+g_.x_),x_Symbol] :=
  1/g*Subst[Int[(a+b*Log[1+c*e*x/g])/x,x],x,f+g*x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[g+c*(e*f-d*g),0]
```

$$\text{3: } \int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{f + g x} dx \text{ when } e f - d g \neq 0$$

Derivation: Integration by parts

$$\text{Basis: } \frac{1}{f + g x} == \frac{1}{g} \partial_x \operatorname{Log} \left[\frac{e (f + g x)}{e f - d g} \right]$$

Rule: If $e f - d g \neq 0$, then

$$\int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{f + g x} dx \rightarrow \frac{\operatorname{Log} \left[\frac{e (f + g x)}{e f - d g} \right] (a + b \operatorname{Log}[c (d + e x)^n])}{g} - \frac{b e n}{g} \int \frac{\operatorname{Log} \left[\frac{e (f + g x)}{e f - d g} \right]}{d + e x} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d+_e_.x_)^n_.])/(f_.+g_.x_),x_Symbol] :=
  Log[e*(f+g*x)/(e*f-d*g)]*(a+b*Log[c*(d+e*x)^n])/g - b*e*n/g*Int[Log[(e*(f+g*x))/(e*f-d*g)]/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0]
```

$$\mathbf{2:} \int (f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n]) dx \text{ when } e f - d g \neq 0 \wedge q \neq -1$$

Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'

Derivation: Integration by parts

$$\text{Basis: } (f + g x)^q = \partial_x \frac{(f + g x)^{q+1}}{g (q+1)}$$

Rule: If $e f - d g \neq 0 \wedge q \neq -1$, then

$$\int (f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n]) dx \rightarrow \frac{(f + g x)^{q+1} (a + b \operatorname{Log}[c (d + e x)^n])}{g (q+1)} - \frac{b e n}{g (q+1)} \int \frac{(f + g x)^{q+1}}{d + e x} dx$$

Program code:

```
Int[(f_.+g_.**x_)^q_.*(a_.+b_.*Log[c_.*(d+_e_.**x_)^n_.]),x_Symbol] :=
  (f+g*x)^(q+1)*(a+b*Log[c*(d+e*x)^n])/(g*(q+1)) -
  b*e*n/(g*(q+1))*Int[(f+g*x)^(q+1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n,q},x] && NeQ[e*f-d*g,0] && NeQ[q,-1]
```

$$\mathbf{2:} \int \frac{(a + b \operatorname{Log}[c (d + e x)^n])^p}{f + g x} dx \text{ when } e f - d g \neq 0 \wedge p - 1 \in \mathbb{Z}^+$$

Derivation: Integration by parts

$$\text{Basis: } \frac{1}{f + g x} = \frac{1}{g} \partial_x \operatorname{Log} \left[\frac{e (f + g x)}{e f - d g} \right]$$

$$\text{Basis: } \partial_x (a + b \operatorname{Log}[c (d + e x)^n])^p = \frac{b e n p (a + b \operatorname{Log}[c (d + e x)^n])^{p-1}}{d + e x}$$

Rule: If $e f - d g \neq 0 \wedge p - 1 \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \operatorname{Log}[c (d + e x)^n])^p}{f + g x} dx \rightarrow \frac{\operatorname{Log}\left[\frac{e (f + g x)}{e f - d g}\right] (a + b \operatorname{Log}[c (d + e x)^n])^p}{g} - \frac{b e n p}{g} \int \frac{\operatorname{Log}\left[\frac{e (f + g x)}{e f - d g}\right] (a + b \operatorname{Log}[c (d + e x)^n])^{p-1}}{d + e x} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_/(f_.+g_.x_),x_Symbol] :=
  Log[e*(f+g*x)/(e*f-d*g)]*(a+b*Log[c*(d+e*x)^n])^p/g -
  b*e*n*p/g*Int[Log[(e*(f+g*x))/(e*f-d*g)]*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && IGtQ[p,1]
```

3: $\int \frac{(a + b \operatorname{Log}[c (d + e x)^n])^p}{(f + g x)^2} dx$ when $e f - d g \neq 0 \wedge p > 0$

Derivation: Integration by parts

Basis: $\frac{1}{(f + g x)^2} = \partial_x \frac{d + e x}{(e f - d g) (f + g x)}$

Rule: If $e f - d g \neq 0 \wedge p > 0$, then

$$\int \frac{(a + b \operatorname{Log}[c (d + e x)^n])^p}{(f + g x)^2} dx \rightarrow \frac{(d + e x) (a + b \operatorname{Log}[c (d + e x)^n])^p}{(e f - d g) (f + g x)} - \frac{b e n p}{e f - d g} \int \frac{(a + b \operatorname{Log}[c (d + e x)^n])^{p-1}}{f + g x} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_/(f_.+g_.x_)^2,x_Symbol] :=
  (d+e*x)*(a+b*Log[c*(d+e*x)^n])^p/((e*f-d*g)*(f+g*x)) -
  b*e*n*p/(e*f-d*g)*Int[(a+b*Log[c*(d+e*x)^n])^(p-1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && GtQ[p,0]
```

$$\mathbf{4:} \int (f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx \text{ when } e f - d g \neq 0 \wedge p > 0 \wedge q \neq -1$$

Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'

Derivation: Integration by parts

$$\text{Basis: } (f + g x)^q = \partial_x \frac{(f + g x)^{q+1}}{g (q+1)}$$

Rule: If $e f - d g \neq 0 \wedge p > 0 \wedge q \neq -1$, then

$$\int (f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow \frac{(f + g x)^{q+1} (a + b \operatorname{Log}[c (d + e x)^n])^p}{g (q+1)} - \frac{b e n p}{g (q+1)} \int \frac{(f + g x)^{q+1} (a + b \operatorname{Log}[c (d + e x)^n])^{p-1}}{d + e x} dx$$

Program code:

```
Int[(f_.+g_.**x_)^q_.*(a_.+b_.*Log[c_.*(d_.+e_.**x_)^n_.])^p_,x_Symbol] :=
  (f+g*x)^(q+1)*(a+b*Log[c*(d+e*x)^n])^p/(g*(q+1)) -
  b*e*n*p/(g*(q+1))*Int[(f+g*x)^(q+1)*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n,q},x] && NeQ[e*f-d*g,0] && GtQ[p,0] && NeQ[q,-1] && IntegersQ[2*p,2*q] &&
  (Not[IGtQ[q,0]] || EqQ[p,2] && NeQ[q,1])
```

$$2. \int (f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx \text{ when } e f - d g \neq 0 \wedge p < 0$$

$$\mathbf{1:} \int \frac{(f + g x)^q}{a + b \operatorname{Log}[c (d + e x)^n]} dx \text{ when } e f - d g \neq 0 \wedge q \in \mathbb{Z}^+$$

Derivation: Algebraic expansion

Note: ExpandIntegrant expresses $(f + g x)^q$ as a polynomial in $d + e x$ so the above rule for when $e f - d g = 0$ will apply.

Rule: If $e f - d g \neq 0 \wedge q \in \mathbb{Z}^+$, then

$$\int \frac{(f+g x)^q}{a+b \log [c (d+e x)^n]} dx \rightarrow \int \text{ExpandIntegrand} \left[\frac{(f+g x)^q}{a+b \log [c (d+e x)^n]}, x \right] dx$$

Program code:

```
Int[(f_.+g_.**x_)^q_./(a_.+b_.*Log[c_.*(d_.+e_.**x_)^n_.]),x_Symbol] :=
  Int[ExpandIntegrand[(f+g*x)^q/(a+b*Log[c*(d+e*x)^n]),x],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && IGtQ[q,0]
```

2: $\int (f+g x)^q (a+b \log [c (d+e x)^n])^p dx$ when $e f - d g \neq 0 \wedge p < -1 \wedge q > 0$

Rule: If $e f - d g \neq 0 \wedge p < -1 \wedge q > 0$, then

$$\int (f+g x)^q (a+b \log [c (d+e x)^n])^p dx \rightarrow$$

$$\frac{(d+e x) (f+g x)^q (a+b \log [c (d+e x)^n])^{p+1}}{b e n (p+1)} +$$

$$\frac{q (e f - d g)}{b e n (p+1)} \int (f+g x)^{q-1} (a+b \log [c (d+e x)^n])^{p+1} dx - \frac{q+1}{b n (p+1)} \int (f+g x)^q (a+b \log [c (d+e x)^n])^{p+1} dx$$

Program code:

```
Int[(f_.+g_.**x_)^q_.*(a_.+b_.*Log[c_.*(d_.+e_.**x_)^n_.])^p_,x_Symbol] :=
  (d+e*x)*(f+g*x)^q*(a+b*Log[c*(d+e*x)^n])^(p+1)/(b*e*n*(p+1)) +
  q*(e*f-d*g)/(b*e*n*(p+1))*Int[(f+g*x)^(q-1)*(a+b*Log[c*(d+e*x)^n])^(p+1),x] -
  (q+1)/(b*n*(p+1))*Int[(f+g*x)^q*(a+b*Log[c*(d+e*x)^n])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && LtQ[p,-1] && GtQ[q,0]
```

$$\mathbf{3:} \int (f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx \text{ when } e f - d g \neq 0 \wedge q \in \mathbb{Z}^+$$

Derivation: Algebraic expansion

Note: `ExpandIntegrand` expresses $(f + g x)^q$ as a polynomial in $d + e x$ so the above rules for when $e f - d g = 0$ will apply.

Rule: If $e f - d g \neq 0 \wedge q \in \mathbb{Z}^+$, then

$$\int (f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow \int \operatorname{ExpandIntegrand}[(f + g x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p, x] dx$$

Program code:

```
Int[(f_.+g_.**x_)^q_.*(a_.+b_.*Log[c_.*(d_.+e_.**x_)^n_.])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(f+g*x)^q*(a+b*Log[c*(d+e*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && IGtQ[q,0]
```

$$2. \int \frac{a + b \operatorname{Log}\left[\frac{c}{d+e x}\right]}{f + g x^2} dx \text{ when } e^2 f + d^2 g == 0 \wedge \frac{c}{2d} > 0$$

$$1: \int \frac{\operatorname{Log}\left[\frac{2d}{d+e x}\right]}{f + g x^2} dx \text{ when } e^2 f + d^2 g == 0$$

Derivation: Integration by substitution

■ Basis: If $e^2 f + d^2 g == 0$, then $\frac{F\left[\frac{1}{d+e x}\right]}{f+g x^2} == -\frac{e}{g} \operatorname{Subst}\left[\frac{F[x]}{1-2 d x}, x, \frac{1}{d+e x}\right] \partial_x \frac{1}{d+e x}$

Rule: If $e^2 f + d^2 g == 0$, then

$$\int \frac{\operatorname{Log}\left[\frac{2d}{d+e x}\right]}{f + g x^2} dx \rightarrow -\frac{e}{g} \operatorname{Subst}\left[\int \frac{\operatorname{Log}[2 d x]}{1-2 d x} dx, x, \frac{1}{d+e x}\right]$$

Program code:

```
Int[Log[c_/(d_+e_.*x_)]/(f_+g_.*x_^2),x_Symbol] :=
  -e/g*Subst[Int[Log[2*d*x]/(1-2*d*x),x],x,1/(d+e*x)] /;
FreeQ[{c,d,e,f,g},x] && EqQ[c,2*d] && EqQ[e^2*f+d^2*g,0]
```

$$2: \int \frac{a + b \operatorname{Log}\left[\frac{c}{d+e x}\right]}{f + g x^2} dx \text{ when } e^2 f + d^2 g == 0 \wedge \frac{c}{2d} > 0$$

Derivation: Algebraic expansion

Basis: If $\frac{c}{2d} > 0$, then $\operatorname{Log}\left[\frac{c}{d+e x}\right] == \operatorname{Log}\left[\frac{c}{2d}\right] \operatorname{Log}\left[\frac{2d}{d+e x}\right]$

Rule: If $e^2 f + d^2 g == 0 \wedge \frac{c}{2d} > 0$, then

$$\int \frac{a + b \operatorname{Log}\left[\frac{c}{d+e x}\right]}{f + g x^2} dx \rightarrow \left(a + b \operatorname{Log}\left[\frac{c}{2 d}\right]\right) \int \frac{1}{f + g x^2} dx + b \int \frac{\operatorname{Log}\left[\frac{2 d}{d+e x}\right]}{f + g x^2} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_./(d_+e_.**x_)])/(f_+g_.**x_^2),x_Symbol] :=
  (a+b*Log[c/(2*d)])*Int[1/(f+g*x^2),x] + b*Int[Log[2*d/(d+e*x)]/(f+g*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e^2*f+d^2*g,0] && GtQ[c/(2*d),0]
```

$$3. \int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{\sqrt{f + g x^2}} dx$$

$$1: \int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{\sqrt{f + g x^2}} dx \text{ when } f > 0$$

Derivation: Integration by parts

$$\text{Basis: } \partial_x (a + b \operatorname{Log}[c (d + e x)^n]) = \frac{b e n}{d + e x}$$

- Note: If $f > 0$, then $\int \frac{1}{\sqrt{f + g x^2}} dx$ involves the inverse sine of a linear function of x , otherwise it involves the inverse tangent of a nonlinear function of x .

- Rule: If $f > 0$, let $u \rightarrow \int \frac{1}{\sqrt{f + g x^2}} dx$, then

$$\int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{\sqrt{f + g x^2}} dx \rightarrow u (a + b \operatorname{Log}[c (d + e x)^n]) - b e n \int \frac{u}{d + e x} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/Sqrt[f_+g_.*x_^2],x_Symbol] :=
  With[{u=IntHide[1/Sqrt[f+g*x^2],x]},
    u*(a+b*Log[c*(d+e*x)^n]) - b*e*n*Int[SimplifyIntegrand[u/(d+e*x),x],x] /;
    FreeQ[{a,b,c,d,e,f,g,n},x] && GtQ[f,0]
```

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/(Sqrt[f1+g1_.*x_] * Sqrt[f2+g2_.*x_]),x_Symbol] :=
  With[{u=IntHide[1/Sqrt[f1+f2+g1*g2*x^2],x]},
    u*(a+b*Log[c*(d+e*x)^n]) - b*e*n*Int[SimplifyIntegrand[u/(d+e*x),x],x] /;
    FreeQ[{a,b,c,d,e,f1,g1,f2,g2,n},x] && EqQ[f2*g1+f1*g2,0] && GtQ[f1,0] && GtQ[f2,0]
```

2: $\int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{\sqrt{f + g x^2}} dx$ when $f \neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1 + \frac{g}{f} x^2}}{\sqrt{f + g x^2}} == 0$

Rule: If $f \neq 0$, then

$$\int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{\sqrt{f + g x^2}} dx \rightarrow \frac{\sqrt{1 + \frac{g}{f} x^2}}{\sqrt{f + g x^2}} \int \frac{a + b \operatorname{Log}[c (d + e x)^n]}{\sqrt{1 + \frac{g}{f} x^2}} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.])/Sqrt[f_+g_.**x_^2],x_Symbol] :=
  Sqrt[1+g/f**x^2]/Sqrt[f+g*x^2]*Int[(a+b*Log[c*(d+e*x)^n])/Sqrt[1+g/f**x^2],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && Not[GtQ[f,0]]
```

```
Int[(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.])/(Sqrt[f1_+g1_.**x_]*Sqrt[f2_+g2_.**x_]),x_Symbol] :=
  Sqrt[1+g1*g2/(f1*f2)**x^2]/(Sqrt[f1+g1*x]*Sqrt[f2+g2*x])*Int[(a+b*Log[c*(d+e*x)^n])/Sqrt[1+g1*g2/(f1*f2)**x^2],x] /;
FreeQ[{a,b,c,d,e,f1,g1,f2,g2,n},x] && EqQ[f2*g1+f1*g2,0]
```

4: $\int (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx$ when $r \in \mathbb{F} \wedge p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x] = k \text{Subst}[x^{k-1} F[x^k], x, x^{1/k}] \partial_x x^{1/k}$

Rule: If $r \in \mathbb{F} \wedge p \in \mathbb{Z}^+$, let $k \rightarrow \text{Denominator}[r]$, then

$$\int (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx \rightarrow k \text{Subst}\left[\int x^{k-1} (f + g x^k)^q (a + b \log[c (d + e x^k)^n])^p dx, x, x^{1/k}\right]$$

Program code:

```
Int[(f_.+g_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*(d+.e_.*x_)^n_.])^p_,x_Symbol] :=
  With[{k=Denominator[r]},
    k*Subst[Int[x^(k-1)*(f+g*x^k)^q*(a+b*Log[c*(d+e*x^k)^n])^p,x],x,x^(1/k)] /;
    FreeQ[{a,b,c,d,e,f,g,n,p,q},x] && FractionQ[r] && IGtQ[p,0]
```

5: $\int (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx$ when $p \in \mathbb{Z}^+ \wedge q \in \mathbb{Z} \wedge (q > 0 \vee (r \in \mathbb{Z} \wedge r \neq 1))$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \wedge q \in \mathbb{Z} \wedge (q > 0 \vee (r \in \mathbb{Z} \wedge r \neq 1))$, then

$$\int (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx \rightarrow \int (a + b \log[c (d + e x)^n])^p \text{ExpandIntegrand}[(f + g x^r)^q, x] dx$$

Program code:

```
Int[(f_.+g_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*(d+.e_.*x_)^n_.])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Log[c*(d+e*x)^n])^p,(f+g*x^r)^q,x],x] /;
  FreeQ[{a,b,c,d,e,f,g,n,r},x] && IGtQ[p,0] && IntegerQ[q] && (GtQ[q,0] || IntegerQ[r] && NeQ[r,1])
```

$$3. \int (f + g x)^q (h + i x)^r (a + b \operatorname{Log}[c (d + e x)^n])^p dx \text{ when } e f - d g == 0$$

$$1: \int \frac{x^m \operatorname{Log}[c (d + e x)]}{f + g x} dx \text{ when } e f - d g == 0 \wedge c d == 1 \wedge m \in \mathbb{Z}$$

Derivation: Algebraic expansion

Rule: If $e f - d g == 0 \wedge c d == 1 \wedge m \in \mathbb{Z}$, then

$$\int \frac{x^m \operatorname{Log}[c (d + e x)]}{f + g x} dx \rightarrow \int \operatorname{Log}[c (d + e x)] \operatorname{ExpandIntegrand}\left[\frac{x^m}{f + g x}, x\right] dx$$

Program code:

```
Int[x^m_.*Log[c_.*(d_+e_.*x_)]/(f_+g_.*x_),x_Symbol] :=
  Int[ExpandIntegrand[Log[c*(d+e*x)],x^m/(f+g*x),x],x] /;
FreeQ[{c,d,e,f,g},x] && EqQ[e*f-d*g,0] && EqQ[c*d,1] && IntegerQ[m]
```


2: $\int (f + g x)^q (h + i x)^r (a + b \operatorname{Log}[c (d + e x)^n])^p dx$ when $e f - d g = 0$

Derivation: Integration by substitution

Basis: $F[x] = \frac{1}{e} \operatorname{Subst}\left[F\left[\frac{x-d}{e}\right], x, d + e x\right] \partial_x (d + e x)$

Rule: If $e f - d g = 0$, then

$$\int (f + g x)^q (h + i x)^r (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow \frac{1}{e} \operatorname{Subst}\left[\int \left(\frac{g x}{e}\right)^q \left(\frac{e h - d i}{e} + \frac{i x}{e}\right)^r (a + b \operatorname{Log}[c x^n])^p dx, x, d + e x\right]$$

Program code:

```
Int[(f_.+g_.x_)^q_.*(h_.+i_.x_)^r_.*(a_.+b_.*Log[c_.*(d_.+e_.x_)^n_.])^p_.,x_Symbol] :=
  1/e*Subst[Int[(g*x/e)^q*((e*h-d*i)/e+i*x/e)^r*(a+b*Log[c*x^n])^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,n,p,q,r},x] && EqQ[e*f-d*g,0] && (IGtQ[p,0] || IGtQ[r,0]) && IntegerQ[2*r]
```

$$4. \int (h x)^m (f + g x^r)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx$$

$$1: \int x^m \left(f + \frac{g}{x}\right)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx \text{ when } m = q \wedge q \in \mathbb{Z}$$

Derivation: Algebraic simplification

Rule: If $m = q \wedge q \in \mathbb{Z}$, then

$$\int x^m \left(f + \frac{g}{x}\right)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow \int (g + f x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx$$

Program code:

```
Int[x_^m_*(f_+g_/x_)^q_*(a_+b_*Log[c_*(d_+e_*x_)^n_])^p_,x_Symbol] :=
  Int[(g+f*x)^q*(a+b*Log[c*(d+e*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q},x] && EqQ[m,q] && IntegerQ[q]
```

2: $\int x^m (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx$ when $m = r - 1 \wedge q \neq -1 \wedge p \in \mathbb{Z}^+$

Derivation: Integration by parts

■

Basis: If $m = r - 1 \wedge q \neq -1$, then $x^m (f + g x^r)^q = \partial_x \frac{(f + g x^r)^{q+1}}{g r (q+1)}$

Rule: If $m = r - 1 \wedge q \neq -1 \wedge p \in \mathbb{Z}^+$, then

$$\int x^m (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx \rightarrow \frac{(f + g x^r)^{q+1} (a + b \log[c (d + e x)^n])^p}{g r (q+1)} - \frac{b e n p}{g r (q+1)} \int \frac{(f + g x^r)^{q+1} (a + b \log[c (d + e x)^n])^{p-1}}{d + e x} dx$$

Program code:

```
Int[x^m_.*(f_.+g_.*x^r_.)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  (f+g*x^r)^(q+1)*(a+b*Log[c*(d+e*x)^n])^p/(g*r*(q+1)) -
  b*e*n*p/(g*r*(q+1))*Int[(f+g*x^r)^(q+1)*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q,r},x] && EqQ[m,r-1] && NeQ[q,-1] && IGtQ[p,0]
```

3: $\int x^m (f + g x^r)^q (a + b \operatorname{Log}[c (d + e x)^n]) dx$ when $m \in \mathbb{Z} \wedge q \in \mathbb{Z} \wedge r \in \mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x (a + b \operatorname{Log}[c (d + e x)^n]) = \frac{b e n}{d + e x}$

Rule: If $m \in \mathbb{Z} \wedge q \in \mathbb{Z} \wedge r \in \mathbb{Z}$, let $u \rightarrow \int x^m (f + g x^r)^q dx$, then

$$\int x^m (f + g x^r)^q (a + b \operatorname{Log}[c (d + e x)^n]) dx \rightarrow u (a + b \operatorname{Log}[c (d + e x)^n]) - b e n \int \frac{u}{d + e x} dx$$

Program code:

```
Int[x^m_.*(f_+g_.*x^r_.)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
  With[{u=IntHide[x^m*(f+g*x^r)^q,x]},
    Dist[(a+b*Log[c*(d+e*x)^n]),u,x] - b*e*n*Int[SimplifyIntegrand[u/(d+e*x),x],x] /;
    InverseFunctionFreeQ[u,x] /;
    FreeQ[{a,b,c,d,e,f,g,m,n,q,r},x] && IntegerQ[m] && IntegerQ[q] && IntegerQ[r]
```

4: $\int x^m (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx$ when $r \in \mathbb{F} \wedge p \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x] = k \text{Subst}[x^{k-1} F[x^k], x, x^{1/k}] \partial_x x^{1/k}$

Rule: If $r \in \mathbb{F} \wedge p \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$, let $k \rightarrow \text{Denominator}[r]$, then

$$\int x^m (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx \rightarrow k \text{Subst}\left[\int x^{k(m+1)-1} (f + g x^k)^q (a + b \log[c (d + e x^k)^n])^p dx, x, x^{1/k}\right]$$

Program code:

```
Int[x^m.*(f.+g.*x^r_)^q.*(a_.+b_.*Log[c.*(d+e.*x_)^n_.])^p_,x_Symbol] :=
  With[{k=Denominator[r]},
    k*Subst[Int[x^(k*(m+1)-1)*(f+g*x^(k*r))^q*(a+b*Log[c*(d+e*x^k)^n])^p,x],x,x^(1/k)] /;
    FreeQ[{a,b,c,d,e,f,g,n,p,q},x] && FractionQ[r] && IGtQ[p,0] && IntegerQ[m]
```

5: $\int (h x)^m (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx$ when $m \in \mathbb{Z} \wedge q \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z} \wedge q \in \mathbb{Z}$, then

$$\int (h x)^m (f + g x^r)^q (a + b \log[c (d + e x)^n])^p dx \rightarrow \int \text{ExpandIntegrand}[(a + b \log[c (d + e x)^n])^p, (h x)^m (f + g x^r)^q, x] dx$$

Program code:

```
Int[(h.*x_)^m.*(f.+g.*x^r_)^q.*(a_.+b_.*Log[c.*(d+e.*x_)^n_.])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Log[c*(d+e*x)^n])^p,(h*x)^m*(f+g*x^r)^q,x],x] /;
  FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q,r},x] && IntegerQ[m] && IntegerQ[q]
```

$$5. \int \text{AF}[x] (a + b \log[c (d + e x)^n])^p dx$$

$$1: \int \text{Poly}[x] (a + b \log[c (d + e x)^n])^p dx$$

Derivation: Algebraic expansion

Rule:

$$\int \text{Poly}[x] (a + b \log[c (d + e x)^n])^p dx \rightarrow \int \text{ExpandIntegrand}[\text{Poly}[x] (a + b \log[c (d + e x)^n])^p, x] dx$$

Program code:

```
Int[Polyx_*(a_.*b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Polyx*(a+b*Log[c*(d+e*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,n,p},x] && PolynomialQ[Polyx,x]
```

2: $\int \text{RF}[x] (a + b \log[c (d + e x)^n])^p dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}$, then

$$\int \text{RF}[x] (a + b \log[c (d + e x)^n])^p dx \rightarrow \int (a + b \log[c (d + e x)^n])^p \text{ExpandIntegrand}[\text{RF}[x], x] dx$$

Program code:

```
Int[RFx_*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*(d+e*x)^n])^p,RFx,x]},
    Int[u,x] /;
    SumQ[u] /;
    FreeQ[{a,b,c,d,e,n},x] && RationalFunctionQ[RFx,x] && IntegerQ[p]
```

```
Int[RFx_*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[RFx*(a+b*Log[c*(d+e*x)^n])^p,x]},
    Int[u,x] /;
    SumQ[u] /;
    FreeQ[{a,b,c,d,e,n},x] && RationalFunctionQ[RFx,x] && IntegerQ[p]
```

U: $\int \text{AF}[x] (a + b \log[c (d + e x)^n])^p dx$

Rule:

$$\int \text{AF}[x] (a + b \log[c (d + e x)^n])^p dx \rightarrow \int \text{AF}[x] (a + b \log[c (d + e x)^n])^p dx$$

Program code:

```
Int[AFx_*(a_.*b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Unintegrable[AFx*(a+b*Log[c*(d+e*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,n,p},x] && AlgebraicFunctionQ[AFx,x,True]
```

N: $\int u^q (a + b \log[c v^n])^p dx$ when $u = f + g x^r \wedge v = d + e x$

Derivation: Algebraic normalization

Rule: If $u = f + g x^r \wedge v = d + e x$, then

$$\int u^q (a + b \log[c v^n])^p dx \rightarrow \int (f + g x)^q (a + b \log[c (d + e x)^n])^p dx$$

Program code:

```
Int[u_^q_.*(a_.*b_.*Log[c_.*v_^n_.])^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^q*(a+b*Log[c*ExpandToSum[v,x]^n])^p,x] /;
FreeQ[{a,b,c,n,p,q},x] && BinomialQ[u,x] && LinearQ[v,x] && Not[BinomialMatchQ[u,x] && LinearMatchQ[v,x]]
```


$$6. \int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n])^p dx$$

$$1: \int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n]) dx$$

Derivation: Integration by parts

$$\text{Basis: } \text{Log}[f x^m] = -\partial_x (x (m - \text{Log}[f x^m]))$$

Rule:

$$\int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n]) dx \rightarrow$$

$$-x (m - \text{Log}[f x^m]) (a + b \text{Log}[c (d + e x)^n]) + b e m n \int \frac{x}{d + e x} dx - b e n \int \frac{x \text{Log}[f x^m]}{d + e x} dx$$

Program code:

```
Int[Log[f_.**x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.]),x_Symbol] :=
  -x*(m-Log[f**x^m])*(a+b*Log[c*(d+e*x)^n]) + b*e*m*n*Int[x/(d+e*x),x] - b*e*n*Int[(x*Log[f*x^m])/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

2: $\int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n])^p dx$ when $p - 1 \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $p - 1 \in \mathbb{Z}^+$, let $u \rightarrow \int (a + b \text{Log}[c (d + e x)^n])^p dx$, then

$$\int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n])^p dx \rightarrow u \text{Log}[f x^m] - m \int \frac{u}{x} dx$$

Program code:

```
Int[Log[f_.**x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.])^p_,x_Symbol] :=
  With[{u=IntHide[(a+b*Log[c*(d+e*x)^n])^p,x]},
    Dist[Log[f*x^m],u,x] - m*Int[Dist[1/x,u,x],x] /;
  FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,1]
```

U: $\int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n])^p dx$

Rule:

$$\int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n])^p dx \rightarrow \int \text{Log}[f x^m] (a + b \text{Log}[c (d + e x)^n])^p dx$$

Program code:

```
Int[Log[f_.**x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.])^p_,x_Symbol] :=
  Unintegrable[Log[f*x^m]*(a+b*Log[c*(d+e*x)^n])^p,x] /;
  FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

$$7. \int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx$$

$$1. \int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n]) dx$$

$$1: \int \frac{\operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])}{x} dx$$

Derivation: Integration by parts

—

$$\text{Basis: } \frac{\operatorname{Log}[f x^m]}{x} == \partial_x \frac{\operatorname{Log}[f x^m]^2}{2 m}$$

Rule:

$$\int \frac{\operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])}{x} dx \rightarrow \frac{\operatorname{Log}[f x^m]^2 (a + b \operatorname{Log}[c (d + e x)^n])}{2 m} - \frac{b e n}{2 m} \int \frac{\operatorname{Log}[f x^m]^2}{d + e x} dx$$

Program code:

```
Int[Log[f_.**x_^m_.]*(a_.+b_.*Log[c_.*(d+_e_.**x_)^n_.])/x_,x_Symbol] :=
  Log[f**x^m]^2*(a+b*Log[c*(d+e*x)^n])/(2*m) - b*e*n/(2*m)*Int[Log[f**x^m]^2/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

$$2: \int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n]) dx \text{ when } q \neq -1$$

Derivation: Integration by parts

—

$$\text{Basis: } (g x)^q \operatorname{Log}[f x^m] == -\frac{1}{g (q+1)} \partial_x \left(\frac{m (g x)^{q+1}}{q+1} - (g x)^{q+1} \operatorname{Log}[f x^m] \right)$$

Rule: If $q \neq -1$, then

$$\int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n]) dx \rightarrow$$

$$-\frac{1}{g (q+1)} \left(\frac{m (g x)^{q+1}}{q+1} - (g x)^{q+1} \operatorname{Log}[f x^m] \right) (a+b \operatorname{Log}[c (d+e x)^n]) + \frac{b e m n}{g (q+1)^2} \int \frac{(g x)^{q+1}}{d+e x} dx - \frac{b e n}{g (q+1)} \int \frac{(g x)^{q+1} \operatorname{Log}[f x^m]}{d+e x} dx$$

Program code:

```
Int[(g_.*x_)^q_.*Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
-1/(g*(q+1))*(m*(g*x)^(q+1)/(q+1)-(g*x)^(q+1)*Log[f*x^m])*(a+b*Log[c*(d+e*x)^n]) +
b*e*m*n/(g*(q+1)^2)*Int[(g*x)^(q+1)/(d+e*x),x] -
b*e*n/(g*(q+1))*Int[(g*x)^(q+1)*Log[f*x^m]/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && NeQ[q,-1]
```

?: $\int \frac{\operatorname{Log}[f x^m] (a+b \operatorname{Log}[c (d+e x)^n])^p}{x} dx$ when $p \in \mathbb{Z}^+$

Derivation: Integration by parts

-

Basis: $\frac{\operatorname{Log}[f x^m]}{x} = \partial_x \frac{\operatorname{Log}[f x^m]^2}{2m}$

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{\operatorname{Log}[f x^m] (a+b \operatorname{Log}[c (d+e x)^n])^p}{x} dx \rightarrow \frac{\operatorname{Log}[f x^m]^2 (a+b \operatorname{Log}[c (d+e x)^n])^p}{2m} - \frac{b e n p}{2m} \int \frac{\operatorname{Log}[f x^m]^2 (a+b \operatorname{Log}[c (d+e x)^n])^{p-1}}{d+e x} dx$$

Program code:

```
Int[Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_./x_,x_Symbol] :=
Log[f*x^m]^2*(a+b*Log[c*(d+e*x)^n])^p/(2*m) - b*e*n*p/(2*m)*Int[Log[f*x^m]^2*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

2: $\int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx$ when $p - 1 \in \mathbb{Z}^+ \wedge q \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $p - 1 \in \mathbb{Z}^+ \wedge q \in \mathbb{Z}^+$, let $u \rightarrow \int (g x)^q (a + b \operatorname{Log}[c (d + e x)^n])^p dx$, then

$$\int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow u \operatorname{Log}[f x^m] - m \int \frac{u}{x} dx$$

Program code:

```
Int[(g_.**x_)^q_.**Log[f_.**x_^m_.]*(a_.+b_.**Log[c_.*(d_.+e_.**x_)^n_.])^p_,x_Symbol] :=
  With[{u=IntHide[(g**x)^q*(a+b*Log[c*(d+e*x)^n])^p,x]},
    Dist[Log[f*x^m],u,x] - m*Int[Dist[1/x,u,x],x] /;
  FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && IGtQ[p,1] && IGtQ[q,0]
```

x: $\int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx$ when $p - 1 \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x ((g x)^q \operatorname{Log}[f x^m]) = g m (g x)^{q-1} + g q (g x)^{q-1} \operatorname{Log}[f x^m]$

Rule: If $p - 1 \in \mathbb{Z}^+$, let $u \rightarrow \int (a + b \operatorname{Log}[c (d + e x)^n])^p dx$, then

$$\int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow u (g x)^q \operatorname{Log}[f x^m] - g m \int u (g x)^{q-1} dx - g q \int u (g x)^{q-1} \operatorname{Log}[f x^m] dx$$

Program code:

```
(* Int[(g_.**x_)^q_.*Log[f_.**x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.])^p_,x_Symbol] :=
  With[{u=IntHide[(a+b*Log[c*(d+e*x)^n])^p,x]},
    Dist[(g*x)^q*Log[f*x^m],u,x] - g*m*Int[Dist[(g*x)^(q-1),u,x],x] - g*q*Int[Dist[(g*x)^(q-1)*Log[f*x^m],u,x],x] /;
  FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && IGtQ[p,1] *)
```

U: $\int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx$

Rule:

$$\int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx \rightarrow \int (g x)^q \operatorname{Log}[f x^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx$$

Program code:

```
Int[(g_.**x_)^q_.*Log[f_.**x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.**x_)^n_.])^p_,x_Symbol] :=
  Unintegrable[(g*x)^q*Log[f*x^m]*(a+b*Log[c*(d+e*x)^n])^p,x] /;
  FreeQ[{a,b,c,d,e,f,g,m,n,p,q},x]
```

$$8. \int (a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m])^q dx$$

$$1: \int (a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m]) dx \text{ when } p \in \mathbb{Z}^+$$

Derivation: Integration by parts

$$\text{Basis: } \partial_x ((f + g \operatorname{Log}[h (i + j x)^m]) (a + b \operatorname{Log}[c (d + e x)^n])^p) = \\ \frac{g j m (a + b \operatorname{Log}[c (d + e x)^n])^p}{i + j x} + \frac{b e n p (a + b \operatorname{Log}[c (d + e x)^n])^{-1+p} (f + g \operatorname{Log}[h (i + j x)^m])}{d + e x}$$

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m]) dx \rightarrow \\ x (a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m]) - \\ g j m \int \frac{x (a + b \operatorname{Log}[c (d + e x)^n])^p}{i + j x} dx - b e n p \int \frac{x (a + b \operatorname{Log}[c (d + e x)^n])^{p-1} (f + g \operatorname{Log}[h (i + j x)^m])}{d + e x} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d+.e.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.]),x_Symbol] :=
  x*(a+b*Log[c*(d+e*x)^n]^p*(f+g*Log[h*(i+j*x)^m]) -
  g*j*m*Int[x*(a+b*Log[c*(d+e*x)^n]^p/(i+j*x),x] -
  b*e*n*p*Int[x*(a+b*Log[c*(d+e*x)^n]^(p-1)*(f+g*Log[h*(i+j*x)^m])/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && IGtQ[p,0]
```

$$2: \int \operatorname{Log}[f (g + h x)^m] (a + b \operatorname{Log}[c (d + e x)^n])^p dx \text{ when } e g - d h = 0$$

Derivation: Integration by substitution

Basis: If $e g - d h = 0$, then

$$\operatorname{Log}[f (g + h x)^m] F[d + e x] = \frac{1}{e} \operatorname{Subst}[\operatorname{Log}[f (\frac{g x}{d})^m] F[x], x, d + e x] \partial_x (d + e x)$$

Rule: If $e g - d h = 0$, then

$$\int \text{Log}\left[f (g + h x)^m\right] (a + b \text{Log}[c (d + e x)^n])^p dx \rightarrow \frac{1}{e} \text{Subst}\left[\int \text{Log}\left[f \left(\frac{g x}{d}\right)^m\right] (a + b \text{Log}[c x^n])^p dx, x, d + e x\right]$$

Program code:

```
Int[Log[f_.*(g_.+h_.*x_)^m_.]*(a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.])^p_.,x_Symbol] :=
  1/e*Subst[Int[Log[f*(g*x/d)^m]*(a+b*Log[c*x^n])^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x] && EqQ[e*f-d*g,0]
```

U: $\int (a + b \text{Log}[c (d + e x)^n])^p (f + g \text{Log}[h (i + j x)^m])^q dx$

Rule:

$$\int (a + b \text{Log}[c (d + e x)^n])^p (f + g \text{Log}[h (i + j x)^m])^q dx \rightarrow \int (a + b \text{Log}[c (d + e x)^n])^p (f + g \text{Log}[h (i + j x)^m])^q dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.])^q_.,x_Symbol] :=
  Unintegrable[(a+b*Log[c*(d+e*x)^n])^p*(f+g*Log[h*(i+j*x)^m])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n,p},x]
```


$$9. \int (k + l x)^r (a + b \log[c (d + e x)^n])^p (f + g \log[h (i + j x)^m])^q dx$$

$$1: \int (k + l x)^r (a + b \log[c (d + e x)^n])^p (f + g \log[h (i + j x)^m]) dx \text{ when } e k - d l = 0$$

Derivation: Integration by substitution

Basis: If $e k - d l = 0$, then $(k + l x)^r F[x] = \frac{1}{e} \text{Subst}\left[\left(\frac{kx}{d}\right)^r F\left[\frac{x-d}{e}\right], x, d + e x\right] \partial_x (d + e x)$

Rule: If $e k - d l = 0$, then

$$\int (k + l x)^r (a + b \log[c (d + e x)^n])^p (f + g \log[h (i + j x)^m]) dx \rightarrow \frac{1}{e} \text{Subst}\left[\int \left(\frac{kx}{d}\right)^r (a + b \log[c x^n])^p \left(f + g \log\left[h \left(\frac{e i - d j}{e} + \frac{j x}{e}\right)^m\right]\right) dx, x, d + e x\right]$$

Program code:

```
Int[(k_.+l_.*x_)^r_.*(a_.+b_.*Log[c_.*(d+.e_.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.]),x_Symbol] :=
  1/e*Subst[Int[(k*x/d)^r*(a+b*Log[c*x^n])^p*(f+g*Log[h*(e*i-d*j)/e+j*x/e]^m)],x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,l,n,p,r},x] && EqQ[e*k-d*l,0]
```

$$2. \int x^r (a + b \log[c (d + e x)^n])^p (f + g \log[h (i + j x)^m]) dx \text{ when } p \in \mathbb{Z}^+ \wedge r \in \mathbb{Z} \wedge (p = 1 \vee r > 0)$$

$$1. \int \frac{(a + b \log[c (d + e x)^n])^p (f + g \log[h (i + j x)^m])}{x} dx$$

$$1. \int \frac{(a + b \log[c (d + e x)^n]) (f + g \log[h (i + j x)^m])}{x} dx$$

$$1: \int \frac{(a + b \log[c (d + e x)^n]) (f + g \log[h (i + j x)^m])}{x} dx \text{ when } e i - d j = 0$$

Derivation: Integration by parts

Basis: If $e i - d j = 0$, then $\partial_x \left((a + b \log[c (d + e x)^n]) (f + g \log[h (i + j x)^m]) \right) =$

$$\frac{e g m (a + b \log[c (d + e x)^n])}{d + e x} + \frac{b j n (f + g \log[h (i + j x)^m])}{i + j x}$$

Rule: If $e i - d j = 0$, then

$$\int \frac{(a + b \log[c (d + e x)^n]) (f + g \log[h (i + j x)^m])}{x} dx \rightarrow$$

$$\log[x] (a + b \log[c (d + e x)^n]) (f + g \log[h (i + j x)^m]) - e g m \int \frac{\log[x] (a + b \log[c (d + e x)^n])}{d + e x} dx - b j n \int \frac{\log[x] (f + g \log[h (i + j x)^m])}{i + j x} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.])/x_,x_Symbol] :=
  Log[x]*(a+b*Log[c*(d+e*x)^n])*(f+g*Log[h*(i+j*x)^m]) -
  e*g*m*Int[Log[x]*(a+b*Log[c*(d+e*x)^n])/(d+e*x),x] -
  b*j*n*Int[Log[x]*(f+g*Log[h*(i+j*x)^m])/(i+j*x),x]/;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && EqQ[e*i-d*j,0]
```

$$2. \int \frac{(a + b \log[c (d + e x)^n]) (f + g \log[h (i + j x)^m])}{x} dx \text{ when } e i - d j \neq 0$$

$$1. \int \frac{\log[c (d + e x)^n] \log[h (i + j x)^m]}{x} dx \text{ when } e i - d j \neq 0$$

$$1: \int \frac{\log[d + e x] \log[i + j x]}{x} dx \text{ when } e i - d j \neq 0$$

Derivation: Integration by parts and ???

Rule: If $b c - a d \neq 0$, then

$$\int \frac{\log[a + b x] \log[c + d x]}{x} dx \rightarrow \log\left[-\frac{b x}{a}\right] \log[a + b x] \log[c + d x] - \int \left(\frac{d \log\left[-\frac{b x}{a}\right] \log[a + b x]}{c + d x} + \frac{b \log\left[-\frac{b x}{a}\right] \log[c + d x]}{a + b x} \right) dx$$

$$\begin{aligned}
& \rightarrow \log\left[-\frac{bx}{a}\right] \log[a+bx] \log[c+dx] - d \left(\log\left[-\frac{bx}{a}\right] - \log\left[-\frac{dx}{c}\right] \right) \int \frac{\log[a+bx] + \log\left[\frac{a(c+dx)}{c(a+bx)}\right]}{c+dx} dx - \\
& (bc-ad) \int \frac{\log\left[-\frac{bx}{a}\right] \log\left[\frac{a(c+dx)}{c(a+bx)}\right]}{(a+bx)(c+dx)} dx - b \int \frac{\log\left[-\frac{bx}{a}\right] \left(\log[c+dx] - \log\left[\frac{a(c+dx)}{c(a+bx)}\right] \right)}{a+bx} dx - d \int \frac{\log\left[-\frac{dx}{c}\right] \left(\log[a+bx] + \log\left[\frac{a(c+dx)}{c(a+bx)}\right] \right)}{c+dx} dx \\
& \rightarrow \log\left[-\frac{bx}{a}\right] \log[a+bx] \log[c+dx] - \frac{1}{2} \left(\log\left[-\frac{bx}{a}\right] - \log\left[-\frac{dx}{c}\right] \right) \left(\log[a+bx] + \log\left[\frac{a(c+dx)}{c(a+bx)}\right] \right)^2 + \\
& \frac{1}{2} \left(\log\left[-\frac{bx}{a}\right] - \log\left[-\frac{(bc-ad)x}{a(c+dx)}\right] + \log\left[\frac{bc-ad}{b(c+dx)}\right] \right) \log\left[\frac{a(c+dx)}{c(a+bx)}\right]^2 + \\
& \left(\log[c+dx] - \log\left[\frac{a(c+dx)}{c(a+bx)}\right] \right) \text{PolyLog}\left[2, 1 + \frac{bx}{a}\right] + \left(\log[a+bx] + \log\left[\frac{a(c+dx)}{c(a+bx)}\right] \right) \text{PolyLog}\left[2, 1 + \frac{dx}{c}\right] - \\
& \log\left[\frac{a(c+dx)}{c(a+bx)}\right] \text{PolyLog}\left[2, \frac{d(a+bx)}{b(c+dx)}\right] + \log\left[\frac{a(c+dx)}{c(a+bx)}\right] \text{PolyLog}\left[2, \frac{c(a+bx)}{a(c+dx)}\right] - \\
& \text{PolyLog}\left[3, 1 + \frac{bx}{a}\right] - \text{PolyLog}\left[3, 1 + \frac{dx}{c}\right] - \text{PolyLog}\left[3, \frac{d(a+bx)}{b(c+dx)}\right] + \text{PolyLog}\left[3, \frac{c(a+bx)}{a(c+dx)}\right]
\end{aligned}$$

Program code:

```

Int[Log[a_+b_.**x_]*Log[c_+d_.**x_]/x_,x_Symbol] :=
  Log[-b*x/a]*Log[a+b*x]*Log[c+d*x] -
  1/2*(Log[-b*x/a]-Log[-d*x/c])*(Log[a+b*x]+Log[a*(c+d*x)/(c*(a+b*x))])^2 +
  1/2*(Log[-b*x/a]-Log[-(b*c-a*d)*x/(a*(c+d*x))]+Log[(b*c-a*d)/(b*(c+d*x))])*Log[a*(c+d*x)/(c*(a+b*x))]^2 +
  (Log[c+d*x]-Log[a*(c+d*x)/(c*(a+b*x))])*PolyLog[2,1+b*x/a] +
  (Log[a+b*x]+Log[a*(c+d*x)/(c*(a+b*x))])*PolyLog[2,1+d*x/c] -
  Log[a*(c+d*x)/(c*(a+b*x))]*PolyLog[2,d*(a+b*x)/(b*(c+d*x))] +
  Log[a*(c+d*x)/(c*(a+b*x))]*PolyLog[2,c*(a+b*x)/(a*(c+d*x))] -
  PolyLog[3,1+b*x/a] - PolyLog[3,1+d*x/c] - PolyLog[3,d*(a+b*x)/(b*(c+d*x))] + PolyLog[3,c*(a+b*x)/(a*(c+d*x))]/;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]

```

```

Int[Log[v_]*Log[w_]/x_,x_Symbol] :=
  Int[Log[ExpandToSum[v,x]]*Log[ExpandToSum[w,x]]/x,x] /;
LinearQ[{v,w},x] && Not[LinearMatchQ[{v,w},x]]

```

$$2: \int \frac{\log[c (d+e x)^n] \log[h (i+j x)^m]}{x} dx \text{ when } e i - d j \neq 0$$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\partial_x (m \log[i+j x] - \log[h (i+j x)^m]) = 0$

Rule: If $e i - d j \neq 0$, then

$$\int \frac{\log[c (d+e x)^n] \log[h (i+j x)^m]}{x} dx \rightarrow m \int \frac{\log[i+j x] \log[c (d+e x)^n]}{x} dx - (m \log[i+j x] - \log[h (i+j x)^m]) \int \frac{\log[c (d+e x)^n]}{x} dx$$

```
Int[Log[c_.*(d_+e_.*x_)^n_.]*Log[h_.*(i_+j_.*x_)^m_.]/x_,x_Symbol] :=
  m*Int[Log[i+j*x]*Log[c*(d+e*x)^n]/x,x] - (m*Log[i+j*x]-Log[h*(i+j*x)^m])*Int[Log[c*(d+e*x)^n]/x,x] /;
FreeQ[{c,d,e,h,i,j,m,n},x] && NeQ[e*i-d*j,0] && NeQ[i+j*x,h*(i+j*x)^m]
```

$$2: \int \frac{(a+b \log[c (d+e x)^n]) (f+g \log[h (i+j x)^m])}{x} dx \text{ when } e g - d h \neq 0$$

Derivation: Algebraic expansion

Rule: If $e i - d j \neq 0$, then

$$\int \frac{(a+b \log[c (d+e x)^n]) (f+g \log[h (i+j x)^m])}{x} dx \rightarrow f \int \frac{a+b \log[c (d+e x)^n]}{x} dx + g \int \frac{\log[h (i+j x)^m] (a+b \log[c (d+e x)^n])}{x} dx$$

Program code:

```
Int[(a_+b_.*Log[c_.*(d_+e_.*x_)^n_.])*(f_+g_.*Log[h_.*(i_+j_.*x_)^m_.])/x_,x_Symbol] :=
  f*Int[(a+b*Log[c*(d+e*x)^n])/x,x] + g*Int[Log[h*(i+j*x)^m]*(a+b*Log[c*(d+e*x)^n])/x,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && NeQ[e*i-d*j,0]
```

$$2: \int x^r (a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m]) dx \text{ when } p \in \mathbb{Z}^+ \wedge r \in \mathbb{Z} \wedge (p = 1 \vee r > 0) \wedge r \neq -1$$

Derivation: Integration by parts

$$\text{Basis: } \partial_x ((a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m])) =$$

$$\frac{g j m (a + b \operatorname{Log}[c (d + e x)^n])^p}{i + j x} + \frac{b e n p (a + b \operatorname{Log}[c (d + e x)^n])^{-1+p} (f + g \operatorname{Log}[h (i + j x)^m])}{d + e x}$$

Rule: If $p \in \mathbb{Z}^+ \wedge r \in \mathbb{Z} \wedge (p = 1 \vee r > 0) \wedge r \neq -1$, then

$$\begin{aligned} & \int x^r (a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m]) dx \rightarrow \\ & \frac{x^{r+1} (a + b \operatorname{Log}[c (d + e x)^n])^p (f + g \operatorname{Log}[h (i + j x)^m])}{r + 1} - \\ & \frac{g j m}{r + 1} \int \frac{x^{r+1} (a + b \operatorname{Log}[c (d + e x)^n])^p}{i + j x} dx - \frac{b e n p}{r + 1} \int \frac{x^{r+1} (a + b \operatorname{Log}[c (d + e x)^n])^{p-1} (f + g \operatorname{Log}[h (i + j x)^m])}{d + e x} dx \end{aligned}$$

Program code:

```
Int[x^r_.*(a_.*b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.*(f_.*g_.*Log[h_.*(i_+j_.*x_)^m_.]),x_Symbol] :=
  x^(r+1)*(a+b*Log[c*(d+e*x)^n])^p*(f+g*Log[h*(i+j*x)^m])/(r+1) -
  g*j*m/(r+1)*Int[x^(r+1)*(a+b*Log[c*(d+e*x)^n])^p/(i+j*x),x] -
  b*e*n*p/(r+1)*Int[x^(r+1)*(a+b*Log[c*(d+e*x)^n])^(p-1)*(f+g*Log[h*(i+j*x)^m])/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && IGtQ[p,0] && IntegerQ[r] && (EqQ[p,1] || GtQ[r,0]) && NeQ[r,-1]
```

$$3: \int (k + l x)^r (a + b \operatorname{Log}[c (d + e x)^n]) (f + g \operatorname{Log}[h (i + j x)^m]) dx \text{ when } r \in \mathbb{Z}$$

Derivation: Integration by substitution

Rule: If $r \in \mathbb{Z}$, then

$$\int (k + l x)^r (a + b \operatorname{Log}[c (d + e x)^n]) (f + g \operatorname{Log}[h (i + j x)^m]) dx \rightarrow$$

$$\frac{1}{l} \text{Subst} \left[\int x^r \left(a + b \log \left[c \left(-\frac{ek-dl}{l} + \frac{ex}{l} \right)^n \right] \right) \left(f + g \log \left[h \left(-\frac{jk-il}{l} + \frac{jx}{l} \right)^m \right] \right) dx, x, k+lx \right]$$

Program code:

```
Int[(k+l*x_)^r.*(a_.+b_.*Log[c_.*(d+e.*x_)^n_.])*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.]),x_Symbol] :=
  1/l*Subst[Int[x^r*(a+b*Log[c*(-(e*k-d*l)/l+e*x/l)^n])*(f+g*Log[h*(-(j*k-i*l)/l+j*x/l)^m]),x],x,k+l*x]/;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,l,m,n},x] && IntegerQ[r]
```

U: $\int (k+lx)^r (a+b \log[c(dx+e)^n])^p (f+g \log[h(i+jx)^m])^q dx$

Rule:

$$\int (k+lx)^r (a+b \log[c(dx+e)^n])^p (f+g \log[h(i+jx)^m])^q dx \rightarrow \int (k+lx)^r (a+b \log[c(dx+e)^n])^p (f+g \log[h(i+jx)^m])^q dx$$

Program code:

```
Int[(k_.+l_.*x_)^r.*(a_.+b_.*Log[c_.*(d+e.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.])^q_. ,x_Symbol] :=
  Unintegrable[(k+l*x)^r*(a+b*Log[c*(d+e*x)^n])^p*(f+g*Log[h*(i+j*x)^m])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,l,m,n,p,q,r},x]
```

10: $\int \frac{\text{PolyLog}[k, h+ix] (a+b \log[c(dx+e)^n])^p}{f+gx} dx$ when $ef-dg=0 \wedge gh-fi=0 \wedge p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] = \frac{1}{e} \text{Subst} \left[F \left[\frac{x-d}{e} \right], x, d+ex \right] \partial_x (d+ex)$

Rule: If $ef-dg=0 \wedge gh-fi=0 \wedge p \in \mathbb{Z}^+$, then

$$\int \frac{\text{PolyLog}[k, h + i x] (a + b \log[c (d + e x)^n])^p}{f + g x} dx \rightarrow \frac{1}{g} \text{Subst} \left[\int \frac{\text{PolyLog}[k, \frac{hx}{d}] (a + b \log[c x^n])^p}{x} dx, x, d + e x \right]$$

Program code:

```
Int[PolyLog[k_, h_+i_.x_]*(a_.+b_.Log[c_.*(d_+e_.x_)^n_.])^p_./(f_+g_.x_), x_Symbol] :=
  1/g*Subst[Int[PolyLog[k, hx/d]*(a+b*Log[c*x^n])^p/x, x], x, d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,k,n}, x] && EqQ[e*f-d*g, 0] && EqQ[g*h-f*i, 0] && IGtQ[p, 0]
```

11: $\int P_x F[f(g+hx)] (a+b \log[c(d+ex)^n]) dx$ when $F \in \{\text{ArcSin}, \text{ArcCos}, \text{ArcTan}, \text{ArcCot}, \text{ArcSinh}, \text{ArcCosh}, \text{ArcTanh}, \text{ArcCoth}\}$

Derivation: Integration by parts

Basis: $\partial_x (a + b \log[c(d+ex)^n]) = \frac{ben}{d+ex}$

Note: If $F \in \{\text{ArcSin}, \text{ArcCos}, \text{ArcTan}, \text{ArcCot}, \text{ArcSinh}, \text{ArcCosh}, \text{ArcTanh}, \text{ArcCoth}\}$, the terms of the antiderivative of $\frac{P_x F[f(g+hx)] dx}{d+ex}$ will be integrable.

Rule: If $F \in \{\text{ArcSin}, \text{ArcCos}, \text{ArcTan}, \text{ArcCot}, \text{ArcSinh}, \text{ArcCosh}, \text{ArcTanh}, \text{ArcCoth}\}$, let $u \rightarrow \int P_x F[f(g+hx)] dx$, then

$$\int P_x F[f(g+hx)] (a+b \log[c(d+ex)^n]) dx \rightarrow u (a+b \log[c(d+ex)^n]) - ben \int \frac{u}{d+ex} dx$$

Program code:

```
Int[Px_.*F_[f_.*(g_.+h_.x_)]*(a_.+b_.Log[c_.*(d_+e_.x_)^n_.]), x_Symbol] :=
  With[{u=IntHide[Px*F[f*(g+hx)], x]},
    Dist[(a+b*Log[c*(d+e*x)^n]), u, x] - b*e*n*Int[SimplifyIntegrand[u/(d+e*x), x], x] /;
  FreeQ[{a,b,c,d,e,f,g,h,n}, x] && PolynomialQ[Px, x] &&
  MemberQ[{ArcSin, ArcCos, ArcTan, ArcCot, ArcSinh, ArcCosh, ArcTanh, ArcCoth}, F]
```

N: $\int u(a+b \log[c v^n])^p dx$ when $v = d+e x$

Derivation: Algebraic normalization

Rule: If $v = d+e x$, then

$$\int u(a+b \log[c v^n])^p dx \rightarrow \int u(a+b \log[c (d+e x)^n])^p dx$$

Program code:

```
Int[u_.*(a_.*b_.*Log[c_.*v_^n_.])^p_.,x_Symbol] :=
  Int[u*(a+b*Log[c*ExpandToSum[v,x]^n])^p,x] /;
FreeQ[{a,b,c,n,p},x] && LinearQ[v,x] && Not[LinearMatchQ[v,x]] && Not[EqQ[n,1] && MatchQ[c*v,e_.*(f_+g_.*x)] /; FreeQ[{e,f,g},x]]]
```


Rules for integrands of the form $u (a + b \operatorname{Log}[c (d (e + f x)^m)^n])^p$

S: $\int u (a + b \operatorname{Log}[c (d (e + f x)^m)^n])^p dx$ when $n \notin \mathbb{Z} \wedge \neg (d \neq 1 \wedge m \neq 1)$

Derivation: Integration by substitution

Rule: If $n \notin \mathbb{Z} \wedge \neg (d \neq 1 \wedge m \neq 1)$, then

$$\int u (a + b \operatorname{Log}[c (d (e + f x)^m)^n])^p dx \rightarrow \operatorname{Subst}\left[\int u (a + b \operatorname{Log}[c d^n (e + f x)^{mn}])^p dx, c d^n (e + f x)^{mn}, c (d (e + f x)^m)^n\right]$$

Program code:

```
Int[u_.*(a_.*b_.*Log[c_.*(d_.*(e_.*f_.*x_)^m_)^n_])^p_.,x_Symbol] :=
  Subst[Int[u*(a+b*Log[c*d^n*(e+f*x)^(m*n)])^p,x],c*d^n*(e+f*x)^(m*n),c*(d*(e+f*x)^m)^n] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]] && Not[EqQ[d,1] && EqQ[m,1]] &&
IntegralFreeQ[IntHide[u*(a+b*Log[c*d^n*(e+f*x)^(m*n)])^p,x]]
```

U: $\int AF[x] (a + b \operatorname{Log}[c (d (e + f x)^m)^n])^p dx$

Rule:

$$\int AF[x] (a + b \operatorname{Log}[c (d (e + f x)^m)^n])^p dx \rightarrow \int AF[x] (a + b \operatorname{Log}[c (d (e + f x)^m)^n])^p dx$$

Program code:

```
Int[AFx_.*(a_.*b_.*Log[c_.*(d_.*(e_.*f_.*x_)^m_)^n_])^p_.,x_Symbol] :=
  Unintegrable[AFx*(a+b*Log[c*(d*(e+f*x)^m)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && AlgebraicFunctionQ[AFx,x,True]
```

