Rules for integrands of the form $\text{Poly}[x]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p$

1. $\int \text{Poly}[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0$

1: $\int \text{Poly}[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $a+b\,z+c\,z^2==(d+e\,z)\left(\frac{a}{d}+\frac{c\,z}{e}\right)$

Rule 1.2.2.3.2.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$, then

$$\int \text{Poly}[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx\,\rightarrow\,\int \text{Poly}[x^2]\,(d+e\,x^2)^{p+q}\left(\frac{a}{d}+\frac{c\,x^2}{e}\right)^p dx$$

Program code:

```
Int[Poly_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[Poly*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p] &&
  (PolyQ[Poly,x^2] || MatchQ[Poly,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

```
Int[Poly_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[Poly*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,q},x] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p] &&
  (PolyQ[Poly,x^2] || MatchQ[Poly,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

2: $\int \text{Poly}[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $\partial_x\frac{(a+b\,x^2+c\,x^4)^p}{(d+e\,x^2)^p\left(\frac{a}{d}+\frac{c\,x^2}{e}\right)^p}==0$

Basis: If $c d^2 - b d e + a e^2 == 0$, then $\frac{(a+b x^2+c x^4)^p}{(d+e x^2)^p \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p} == \frac{(a+b x^2+c x^4)^{\text{FracPart}[p]}}{(d+e x^2)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^{\text{FracPart}[p]}}$

Rule 1.2.2.3.2.2: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 == 0 \land p \notin \mathbb{Z}$, then

$$\int \text{Poly}[x^2] \, (d + e x^2)^q \, (a + b x^2 + c x^4)^p \, dx \rightarrow \frac{(a + b x^2 + c x^4)^{\text{FracPart}[p]}}{(d + e x^2)^{\text{FracPart}[p]} \left(\frac{a}{d} + \frac{c x^2}{e}\right)^{\text{FracPart}[p]}} \int \text{Poly}[x^2] \, (d + e x^2)^{p+q} \left(\frac{a}{d} + \frac{c x^2}{e}\right)^p dx$$

Program code:

```
Int[Poly_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+c*x^2/e)^FracPart[p])*
    Int[Poly*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] &&
  (PolyQ[Poly,x^2] || MatchQ[Poly,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

```
Int[Poly_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_,x_Symbol] :=
  (a+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+c*x^2/e)^FracPart[p])*
    Int[Poly*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,p,q},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] &&
  (PolyQ[Poly,x^2] || MatchQ[Poly,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

2: $\int \frac{\left(a+b x^2+c x^4\right)^p \left(A+B x^2+C x^4\right)}{d+e x^2} \, dx$ when $b^2-4 a c \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B x^2+C x^4}{d+e x^2} == -\frac{C d-B e-C e x^2}{e^2}+\frac{C d^2-B d e+A e^2}{e^2 \left(d+e x^2\right)}$

Rule 1.2.2.3.6.2.4: If $b^2-4 a c \neq 0$, then

$$\int \frac{\left(a+b x^2+c x^4\right)^p \left(A+B x^2+C x^4\right)}{d+e x^2} \, dx \longrightarrow -\frac{1}{e^2} \int \left(a+b x^2+c x^4\right)^p \left(C d-B e-C e x^2\right) dx+\frac{C d^2-B d e+A e^2}{e^2} \int \frac{\left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_.*Poly4_/(d_+e_.*x_^2),x_Symbol] :=
  With[{A=Coeff[Poly4,x,0],B=Coeff[Poly4,x,2],C=Coeff[Poly4,x,4]},
  -1/e^2*Int[(a+b*x^2+c*x^4)^p*(C*d-B*e-C*e*x^2),x] +
  (C*d^2-B*d*e+A*e^2)/e^2*Int[(a+b*x^2+c*x^4)^p/(d+e*x^2),x]] /;
FreeQ[{a,b,c,d,e,p},x] && PolyQ[Poly4,x^2] && EqQ[Expon[Poly4,x],4] && NeQ[b^2-4*a*c,0]
```

```
Int[(a_+c_.*x_^4)^p_.*Poly4_/(d_+e_.*x_^2),x_Symbol] :=
  With[{A=Coeff[Poly4,x,0],B=Coeff[Poly4,x,2],C=Coeff[Poly4,x,4]},
  -1/e^2*Int[(a+c*x^4)^p*(C*d-B*e-C*e*x^2),x] +
  (C*d^2-B*d*e+A*e^2)/e^2*Int[(a+c*x^4)^p/(d+e*x^2),x]] /;
FreeQ[{a,c,d,e,p},x] && PolyQ[Poly4,x^2] && EqQ[Expon[Poly4,x],4]
```

3: $\int \frac{\text{Poly}\left[x^2\right] \left(d+e x^2\right)^q}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0$

Derivation: Algebraic expansion

Rule 1.2.2.3.6.2.4: If $b^2-4 a c \neq 0 \land q \in \mathbb{Z}^+$, then

$$\int \frac{\text{Poly}[x^2]\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{\text{Poly}[x^2]\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4},\,x\right]dx$$

Program code:

```
Int[Poly_*(d_+e_.*x_^2)^q_./(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[Poly*(d+e*x^2)^q/(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c,d,e,q},x] && PolyQ[Poly,x^2]
```

```
Int[Poly_*(d_+e_.*x_^2)^q_./(a_+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[Poly*(d+e*x^2)^q/(a+c*x^4),x],x] /;
FreeQ[{a,c,d,e,q},x] && PolyQ[Poly,x^2]
```

4. $\int \frac{\text{Poly}[x^2]\,(d+e\,x^2)^q}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \frac{(d+e\,x^2)^q\,(A+B\,x^2+C\,x^4)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,q\in\mathbb{Z}^+$

Rule 1.2.2.3.6.2.4: If $b^2-4\,a\,c\neq 0\,\wedge\,q\in\mathbb{Z}^+$, then

$$\int \frac{(d+e\,x^2)^q\,(A+B\,x^2+C\,x^4)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx\,\longrightarrow$$

$$\frac{C\,x\,(d+e\,x^2)^q\,\sqrt{a+b\,x^2+c\,x^4}}{c\,(2\,q+3)}+$$

$$\frac{1}{c\,(2\,q+3)}\int\Big(\Big((d+e\,x^2)^{q-1}\,\big(A\,c\,d\,(2\,q+3)-a\,C\,d+\big(c\,(B\,d+A\,e)\,(2\,q+3)-C\,(2\,b\,d+a\,e+2\,a\,e\,q)\big)\,x^2+\big(B\,c\,e\,(2\,q+3)-2\,C\,(b\,e-c\,d\,q+b\,e\,q)\big)\,x^4\big)\Big)\Big/\Big(\sqrt{a+b\,x^2+c\,x^4}\Big)\Big)\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*Poly4_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Poly4,x,0],B=Coeff[Poly4,x,2],C=Coeff[Poly4,x,4]},
  C*x*(d+e*x^2)^q*Sqrt[a+b*x^2+c*x^4]/(c*(2*q+3)) +
  1/(c*(2*q+3))*Int[(d+e*x^2)^(q-1)/Sqrt[a+b*x^2+c*x^4]*
     Simp[A*c*d*(2*q+3)-a*C*d+(c*(B*d+A*e)*(2*q+3)-C*(2*b*d+a*e+2*a*e*q))*x^2+(B*c*e*(2*q+3)-2*C*(b*e-c*d*q+b*e*q))*x^4,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Poly4,x^2] && EqQ[Expon[Poly4,x],4] && NeQ[b^2-4*a*c,0] && IGtQ[q,0]
```

```
Int[(d_+e_.*x_^2)^q_*Poly4_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Poly4,x,0],B=Coeff[Poly4,x,2],C=Coeff[Poly4,x,4]},
  C*x*(d+e*x^2)^q*Sqrt[a+c*x^4]/(c*(2*q+3)) +
  1/(c*(2*q+3))*Int[(d+e*x^2)^(q-1)/Sqrt[a+c*x^4]*
     Simp[A*c*d*(2*q+3)-a*C*d+(c*(B*d+A*e)*(2*q+3)-a*C*e*(2*q+1))*x^2+(B*c*e*(2*q+3)+2*c*C*d*q)*x^4,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Poly4,x^2] && EqQ[Expon[Poly4,x],4] && IGtQ[q,0]
```

2: $\int \frac{(d+e x^2)^q (A+B x^2+C x^4)}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land q \in \mathbb{Z}^- - 1$

Rule 1.2.2.3.6.2.4: If $b^2-4 a c \neq 0 \land q \in \mathbb{Z}^- - 1$, then

$$\int \frac{(d+e x^2)^q (A+B x^2+C x^4)}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow$$

$$-\frac{(C d^2-B d e+A e^2)\, x\, (d+e x^2)^{q+1} \sqrt{a+b x^2+c x^4}}{2 d (q+1) (c d^2-b d e+a e^2)} + \frac{1}{2 d (q+1) (c d^2-b d e+a e^2)} \int \frac{(d+e x^2)^{q+1}}{\sqrt{a+b x^2+c x^4}} \cdot$$
$$\big(a d (C d-B e)+A (a e^2 (2 q+3)+2 d (c d-b e) (q+1)) -$$
$$2 ((B d-A e) (b e (q+2)-c d (q+1))-C d (b d+a e (q+1))) x^2+c (C d^2-B d e+A e^2) (2 q+5) x^4\big) \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*Poly4_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Poly4,x,0],B=Coeff[Poly4,x,2],C=Coeff[Poly4,x,4]},
  -(C*d^2-B*d*e+A*e^2)*x*(d+e*x^2)^(q+1)*Sqrt[a+b*x^2+c*x^4]/(2*d*(q+1)*(c*d^2-b*d*e+a*e^2)) +
  1/(2*d*(q+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x^2)^(q+1)/Sqrt[a+b*x^2+c*x^4]*
    Simp[a*d*(C*d-B*e)+A*(a*e^2*(2*q+3)+2*d*(c*d-b*e)*(q+1))-
      2*((B*d-A*e)*(b*e*(q+2)-c*d*(q+1))-C*d*(b*d+a*e*(q+1)))*x^2+
      c*(C*d^2-B*d*e+A*e^2)*(2*q+5)*x^4,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Poly4,x^2] && LeQ[Expon[Poly4,x],4] && NeQ[b^2-4*a*c,0] && ILtQ[q,-1]
```

```
Int[(d_+e_.*x_^2)^q_*Poly4_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[Poly4,x,0],B=Coeff[Poly4,x,2],C=Coeff[Poly4,x,4]},
  -(C*d^2-B*d*e+A*e^2)*x*(d+e*x^2)^(q+1)*Sqrt[a+c*x^4]/(2*d*(q+1)*(c*d^2+a*e^2)) +
  1/(2*d*(q+1)*(c*d^2+a*e^2))*Int[(d+e*x^2)^(q+1)/Sqrt[a+c*x^4]*
    Simp[a*d*(C*d-B*e)+A*(a*e^2*(2*q+3)+2*c*d^2*(q+1))+2*d*(B*c*d-A*c*e+a*C*e)*(q+1)*x^2+c*(C*d^2-B*d*e+A*e^2)*(2*q+5)*x^4,x],x]
FreeQ[{a,c,d,e},x] && PolyQ[Poly4,x^2] && LeQ[Expon[Poly4,x],4] && ILtQ[q,-1]
```

3. $\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$

1: $\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $e f+d g==0 \land c d^2-a e^2==0$

Derivation: Integration by substitution

Basis: If $e f+d g==0 \land c d^2-a e^2==0$, then $\frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}} == f\,\text{Subst}\left[\frac{1}{d-(b d-2 a e) x^2}, x, \frac{x}{\sqrt{a+b x^2+c x^4}}\right] \partial_x \frac{x}{\sqrt{a+b x^2+c x^4}}$

Rule: If $e f+d g==0 \land c d^2-a e^2==0$, then

$$\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \to f\,\text{Subst}\left[\int \frac{1}{d-(b d-2 a e) x^2}\,dx, x, \frac{x}{\sqrt{a+b x^2+c x^4}}\right]$$

Program code:

```
Int[(f_+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[v_]),x_Symbol] :=
  With[{a=Coeff[v,x,0],b=Coeff[v,x,2],c=Coeff[v,x,4]},
  f*Subst[Int[1/(d-(b*d-2*a*e)*x^2),x],x,x/Sqrt[v]] /;
 EqQ[g*d+f*e,0] && EqQ[c*d^2-a*e^2,0]] /;
FreeQ[{d,e,f,g},x] && PolyQ[v,x^2,2]
```

2. $\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c f^2-b f g+a g^2 \neq 0 \land \frac{c}{a}>0$

1: $\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c f^2-b f g+a g^2 \neq 0 \land \frac{c}{a}>0 \land c f^2-a g^2==0$

Rule: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c f^2-b f g+a g^2 \neq 0 \land \frac{c}{a}>0 \land c f^2-a g^2==0$, let $q \to \sqrt{\frac{g}{f}}$, then

$$\int \frac{f+g x^2}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow$$

$$\frac{(e f-d g)\,\text{ArcTan}\left[\frac{\sqrt{-b+\frac{c d}{e}+\frac{a e}{d}}\,x}{\sqrt{a+b x^2+c x^4}}\right]}{2 d e\sqrt{-b+\frac{c d}{e}+\frac{a e}{d}}}+\frac{(e f+d g)\left(f+g x^2\right)\sqrt{\frac{f^2\left(a+b x^2+c x^4\right)}{a\left(f+g x^2\right)^2}}}{4 d e f q\sqrt{a+b x^2+c x^4}}\,\text{EllipticPi}\left[-\frac{(e f-d g)^2}{4 d e f g},\,2\,\text{ArcTan}[q\,x],\,\frac{1}{2}-\frac{b f}{4 a g}\right]$$

Program code:

```
Int[(f_+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[g/f,2]},
  (e*f-d*g)*ArcTan[Sqrt[-b+c*d/e+a*e/d]*x/Sqrt[a+b*x^2+c*x^4]]/(2*d*e*Sqrt[-b+c*d/e+a*e/d]) +
  (e*f+d*g)*(f+g*x^2)*Sqrt[f^2*(a+b*x^2+c*x^4)/(a*(f+g*x^2)^2)]/(4*d*e*f*q*Sqrt[a+b*x^2+c*x^4])*
    EllipticPi[-(e*f-d*g)^2/(4*d*e*f*g),2*ArcTan[q*x],1/2-b*f/(4*a*g)]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*f^2-b*f*g+a*g^2,0] && PosQ[c/a] && EqQ[c*f^2-a
```

```
Int[(f_+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[g/f,2]},
  (e*f-d*g)*ArcTan[Sqrt[c*d/e+a*e/d]*x/Sqrt[a+c*x^4]]/(2*d*e*Sqrt[c*d/e+a*e/d]) +
  (e*f+d*g)*(f+g*x^2)*Sqrt[f^2*(a+c*x^4)/(a*(f+g*x^2)^2)]/(4*d*e*f*q*Sqrt[a+c*x^4])*
    EllipticPi[-(e*f-d*g)^2/(4*d*e*f*g),2*ArcTan[q*x],1/2]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*f^2+a*g^2,0] && PosQ[c/a] && EqQ[c*f^2-a*g^2,0]
```

2: $\int \frac{f+g x^2}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c f^2-b f g+a g^2\neq 0 \land \frac{c}{a}>0 \land c f^2-a g^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{f+g x^2}{d+e x^2} == \frac{g-f q}{e-d q}+\frac{(e f-d g)\left(1+q x^2\right)}{(e-d q)\left(d+e x^2\right)}$

Rule: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c f^2-b f g+a g^2\neq 0 \land \frac{c}{a}>0 \land c f^2-a g^2\neq 0$, let $q\to\sqrt{\frac{c}{a}}$, then

$$\int \frac{f+g x^2}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx \to \frac{g-f q}{e-d q}\int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx+\frac{e f-d g}{e-d q}\int \frac{1+q x^2}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[(f_.+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (g-f*q)/(e-d*q)*Int[1/Sqrt[a+b*x^2+c*x^4],x] +
  (e*f-d*g)/(e-d*q)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
 NeQ[g,f*q]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*f^2-b*f*g+a*g^2,0] && PosQ[c/a] && NeQ[c*f^2-a
```

```
Int[(f_.+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (g-f*q)/(e-d*q)*Int[1/Sqrt[a+c*x^4],x] +
  (e*f-d*g)/(e-d*q)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
 NeQ[g,f*q]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*f^2+a*g^2,0] && PosQ[c/a] && NeQ[c*f^2-a*g^2,0]
```

3: $\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c f^2-b f g+a g^2\neq 0 \land \frac{c}{a}\ngtr 0$

Derivation: Algebraic expansion

Basis: $\frac{f+g x^2}{d+e x^2} == \frac{g}{e}+\frac{e f-d g}{e\,(d+e x^2)}$

Rule: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c f^2-b f g+a g^2\neq 0 \land \frac{c}{a}\ngtr 0$, then

$$\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \to \frac{g}{e}\int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx+\frac{e f-d g}{e}\int \frac{1}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[(f_.+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  g/e*Int[1/Sqrt[a+b*x^2+c*x^4],x] + (e*f-d*g)/e*Int[1/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*f^2-b*f*g+a*g^2,0] && NegQ[c/a]
```

```
Int[(f_.+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  g/e*Int[1/Sqrt[a+c*x^4],x] + (e*f-d*g)/e*Int[1/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*f^2+a*g^2,0] && NegQ[c/a]
```

x: $\int \frac{f+g x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac>0 \land c d^2-b d e+a e^2\neq 0 \land c f^2-b f g+a g^2\neq 0 \land c\nless 0 \land 2 c f-g\left(b-\sqrt{b^2-4ac}\right)\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{f+g x^2}{d+e x^2} == \frac{2 c f-g\,(b-q)}{2 c d-e\,(b-q)}-\frac{(e f-d g)\,(b-q+2 c x^2)}{(2 c d-e\,(b-q))\,(d+e x^2)}$

Rule: If $b^2-4ac>0 \land c d^2-b d e+a e^2\neq 0 \land c f^2-b f g+a g^2\neq 0 \land c\nless 0$, let $q\to\sqrt{b^2-4ac}$, if $2 c d-e\,(b-q)\neq 0 \land 2 c f-g\,(b-q)\neq 0$, then

$$\int \frac{f + g x^2}{\left(d + e x^2\right) \sqrt{a + b x^2 + c x^4}} \, dx \to \frac{2 c f - g \left(b - q\right)}{2 c d - e \left(b - q\right)} \int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx - \frac{e f - d g}{2 c d - e \left(b - q\right)} \int \frac{b - q + 2 c x^2}{\left(d + e x^2\right) \sqrt{a + b x^2 + c x^4}} \, dx$$

Program code:

```
(* Int[(f_.+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (2*c*f-g*(b-q))/(2*c*d-e*(b-q))*Int[1/Sqrt[a+b*x^2+c*x^4],x] -
  (e*f-d*g)/(2*c*d-e*(b-q))*Int[(b-q+2*c*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
 NeQ[2*c*f-g*(b-q),0]] /;
FreeQ[{a,b,c,d,e,f,g},x] && GtQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*f^2-b*f*g+a*g^2,0] && Not[LtQ[c,0]] *)
```

```
(* Int[(f_+g_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  (c*f+g*q)/(c*d+e*q)*Int[1/Sqrt[a+c*x^4],x] + (e*f-d*g)/(c*d+e*q)*Int[(q-c*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
 NeQ[c*f+g*q,0]] /;
FreeQ[{a,c,d,e,f,g},x] && GtQ[-a*c,0] && NeQ[c*d^2+a*e^2,0] && NeQ[c*f^2+a*g^2,0] && Not[LtQ[c,0]] *)
```

U: $\int \text{Poly}\left[x^2\right] \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p \, dx$

Rule:

$$\int \text{Poly}\left[x^2\right] \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p \, dx \to \int \text{Poly}\left[x^2\right] \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p \, dx$$

Program code:

```
Int[Poly_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[Poly*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x] && PolyQ[Poly,x^2]
```

```
Int[Poly_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[Poly*(d+e*x^2)^q*(a+c*x^4)^p,x] /;
FreeQ[{a,c,d,e,p,q},x] && PolyQ[Poly,x^2]
```

Rules for integrands of the form Poly[x] $(d+e\,x)^q\left(a+b\,x^2+c\,x^4\right)^p$

1: $\int \frac{\text{Poly}[x^2]\left(a+b\,x^2+c\,x^4\right)^p}{d+e\,x}\,dx$ when $p+\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e\,x}==\frac{d}{d^2-e^2\,x^2}-\frac{e\,x}{d^2-e^2\,x^2}$

Rule 1.2.2.9.1: If $p+\frac{1}{2}\in\mathbb{Z}$, then

$$\int \frac{\text{Poly}[x^2]\left(a+b\,x^2+c\,x^4\right)^p}{d+e\,x}\,dx\;\longrightarrow\;d\int \frac{\text{Poly}[x^2]\left(a+b\,x^2+c\,x^4\right)^p}{d^2-e^2\,x^2}\,dx-e\int \frac{x\,\text{Poly}[x^2]\left(a+b\,x^2+c\,x^4\right)^p}{d^2-e^2\,x^2}\,dx$$

Program code:

```
Int[Poly_*u_^p_./(d_+e_.*x_),x_Symbol] :=
  d*Int[Poly*u^p/(d^2-e^2*x^2),x] - e*Int[x*Poly*u^p/(d^2-e^2*x^2),x] /;
FreeQ[{d,e},x] && PolyQ[Poly,x^2] && PolyQ[u,x^2,2] && IntegerQ[p+1/2]
```

2. $\int \frac{f+g\,x}{(d+e\,x)\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $e\,f-d\,g\neq 0$

1: $\int \frac{f+g\,x}{(d+e\,x)\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $e\,f-d\,g\neq 0\,\land\,e^2\,f^2+4\,d\,e\,f\,g+d^2\,g^2==0\,\land\,4\,c\,d\,e\,f+2\,c\,d^2\,g-b\,e^2\,g==0\,\land\,4\,a\,e^5\,f+c\,d^5\,g+15\,a\,d\,e^4\,g==0$

Derivation: Integration by substitution

Basis: If $e^2\,f^2+4\,d\,e\,f\,g+d^2\,g^2==0\,\land\,4\,c\,d\,e\,f+2\,c\,d^2\,g-b\,e^2\,g==0\,\land\,4\,a\,e^5\,f+c\,d^5\,g+15\,a\,d\,e^4\,g==0$, then

$$\frac{f+g\,x}{(d+e\,x)\sqrt{a+b\,x^2+c\,x^4}}==-\frac{f^2\,(e\,f+d\,g)}{e}\,\text{Subst}\left[\frac{1}{3\,e\,f^4+6\,d\,f^3\,g-a\,e\,x^2},\,x,\,\frac{(f+g\,x)^2}{\sqrt{a+b\,x^2+c\,x^4}}\right]\partial_x\frac{(f+g\,x)^2}{\sqrt{a+b\,x^2+c\,x^4}}$$

Rule 1.2.2.9.2.1: If $e\,f-d\,g\neq 0\,\land\,e^2\,f^2+4\,d\,e\,f\,g+d^2\,g^2==0\,\land\,4\,c\,d\,e\,f+2\,c\,d^2\,g-b\,e^2\,g==0\,\land\,4\,a\,e^5\,f+c\,d^5\,g+15\,a\,d\,e^4\,g==0$, then

$$\int\frac{f+g\,x}{(d+e\,x)\sqrt{a+b\,x^2+c\,x^4}}\,dx\,\rightarrow\,-\frac{f^2\,(e\,f+d\,g)}{e}\,\text{Subst}\left[\int\frac{1}{3\,e\,f^4+6\,d\,f^3\,g-a\,e\,x^2}\,dx,\,x,\,\frac{(f+g\,x)^2}{\sqrt{a+b\,x^2+c\,x^4}}\right]$$

Program code:

```
Int[(f_+g_.*x_)/((d_+e_.*x_)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  -f^2*(e*f+d*g)/e*Subst[Int[1/(3*e*f^4+6*d*f^3*g-a*e*x^2),x],x,(f+g*x)^2/Sqrt[a+b*x^2+c*x^4]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[e^2*f^2+4*d*e*f*g+d^2*g^2,0] && EqQ[4*c*d*e*f+2*c*d^2*g-b*e^2*g,0] &&
  EqQ[4*a*e^5*f+c*d^5*g+15*a*d*e^4*g,0]
```

2: $\int \frac{f+g\,x}{(d+e\,x)\,\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $e\,f-d\,g\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{f+g\,x}{d+e\,x} == -\frac{(e\,f-d\,g)\,x}{d^2-e^2\,x^2}+\frac{d\,f-e\,g\,x^2}{d^2-e^2\,x^2}$

Rule 1.2.2.9.2.2: If $e\,f-d\,g\neq 0$, then

$$\int \frac{f+g\,x}{(d+e\,x)\,\sqrt{a+b\,x^2+c\,x^4}}\,dx \longrightarrow -(e\,f-d\,g)\int \frac{x}{(d^2-e^2\,x^2)\,\sqrt{a+b\,x^2+c\,x^4}}\,dx+\int \frac{d\,f-e\,g\,x^2}{(d^2-e^2\,x^2)\,\sqrt{a+b\,x^2+c\,x^4}}\,dx$$

Program code:

```
Int[(f_+g_.*x_)/((d_+e_.*x_)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  -(e*f-d*g)*Int[x/((d^2-e^2*x^2)*Sqrt[a+b*x^2+c*x^4]),x] +
  Int[(d*f-e*g*x^2)/((d^2-e^2*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0]
```

```
Int[(f_+g_.*x_)/((d_+e_.*x_)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  -(e*f-d*g)*Int[x/((d^2-e^2*x^2)*Sqrt[a+c*x^4]),x] +
  Int[(d*f-e*g*x^2)/((d^2-e^2*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[e*f-d*g,0]
```