Rules for integrands of the form $u\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q$

0: $\int P_q[x]^m\,\text{Log}[F[x]]\,dx$ when $m\in\mathbb{Z}\,\land\,C==\frac{P_q[x]^m\,(1-F[x])}{\partial_x F[x]}$

Derivation: Integration by substitution

Basis: If $C==\frac{P_q[x]^m\,(1-F[x])}{\partial_x F[x]}$, then $P_q[x]^m\,\text{Log}[F[x]]==C\,\text{Subst}\left[\frac{\text{Log}[x]}{1-x},x,F[x]\right]\partial_x F[x]$

Rule: If $m\in\mathbb{Z}\,\land\,C==\frac{P_q[x]^m\,(1-F[x])}{\partial_x F[x]}$, then

$$\int P_q[x]^m\,\text{Log}[F[x]]\,dx\,\longrightarrow\,C\,\text{Subst}\left[\int\frac{\text{Log}[x]}{1-x}\,dx,x,u\right]\,\longrightarrow\,C\,\text{PolyLog}[2,1-u]$$

Program code:

```
Int[Pq_^m_.*Log[u_],x_Symbol] :=
  With[{C=FullSimplify[Pq^m*(1-u)/D[u,x]]},
  C*PolyLog[2,1-u] /;
 FreeQ[C,x]] /;
IntegerQ[m] && PolyQ[Pq,x] && RationalFunctionQ[u,x] && LeQ[RationalFunctionExponents[u,x][[2]],Expon[Pq,x]]
```

1. $\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx$

1: $\int \operatorname{Log}[c (d + e x^n)^p] \, dx$

Derivation: Integration by parts

Rule:

$$\int \operatorname{Log}[c (d + e x^n)^p] \, dx \longrightarrow x \operatorname{Log}[c (d + e x^n)^p] - e\, n\, p \int \frac{x^n}{d + e x^n} \, dx$$

Program code:

```
Int[Log[c_.*(d_+e_.*x_^n_)^p_.],x_Symbol] :=
  x*Log[c*(d+e*x^n)^p] - e*n*p*Int[x^n/(d+e*x^n),x] /;
FreeQ[{c,d,e,n,p},x]
```

2. $\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx$ when $q \in \mathbb{Z}^+ \wedge (q == 1 \vee n \in \mathbb{Z})$

1: $\int \left(a + b \operatorname{Log}\left[c \left(d + \frac{e}{x}\right)^p\right]\right)^q dx$ when $q \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $q \in \mathbb{Z}^+$, then

$$\int \left(a + b \operatorname{Log}\left[c \left(d + \frac{e}{x}\right)^p\right]\right)^q dx \longrightarrow \frac{(e + d x) \left(a + b \operatorname{Log}\left[c \left(d + \frac{e}{x}\right)^p\right]\right)^q}{d} + \frac{b e p q}{d} \int \frac{\left(a + b \operatorname{Log}\left[c \left(d + \frac{e}{x}\right)^p\right]\right)^{q-1}}{x} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_./x_)^p_.])^q_,x_Symbol] :=
  (e+d*x)*(a+b*Log[c*(d+e/x)^p])^q/d + b*e*p*q/d*Int[(a+b*Log[c*(d+e/x)^p])^(q-1)/x,x] /;
FreeQ[{a,b,c,d,e,p},x] && IGtQ[q,0]
```

2: $\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx$ when $q \in \mathbb{Z}^+ \wedge (q == 1 \vee n \in \mathbb{Z})$

Derivation: Integration by parts

Rule: If $q \in \mathbb{Z}^+ \wedge (q == 1 \vee n \in \mathbb{Z})$, then

$$\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q dx \longrightarrow x (a + b \operatorname{Log}[c (d + e x^n)^p])^q - b e n p q \int \frac{x^n (a + b \operatorname{Log}[c (d + e x^n)^p])^{q-1}}{d + e x^n} dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_,x_Symbol] :=
  x*(a+b*Log[c*(d+e*x^n)^p])^q - b*e*n*p*q*Int[x^n*(a+b*Log[c*(d+e*x^n)^p])^(q-1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && IGtQ[q,0] && (EqQ[q,1] || IntegerQ[n])
```

x: $\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx$ when $-1 < n < 1 \land (n > 0 \lor q \in \mathbb{Z}^+)$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x^n] = k \operatorname{Subst}[x^{k-1} F[x^{k n}], x, x^{1/k}] \, \partial_x x^{1/k}$

Rule: If $-1 < n < 1 \land (n > 0 \lor q \in \mathbb{Z}^+)$, let $k \to \operatorname{Denominator}[n]$, then

$$\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx \longrightarrow k \operatorname{Subst}\left[\int x^{k-1} (a + b \operatorname{Log}[c (d + e x^{k n})^p])^q \, dx, x, x^{1/k}\right]$$

Program code:

```
(* Int[(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*Log[c*(d+e*x^(k*n))^p])^q,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e,p,q},x] && LtQ[-1,n,1] && (GtQ[n,0] || IGtQ[q,0]) *)
```

3: $\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx$ when $n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x^n] = k \operatorname{Subst}[x^{k-1} F[x^{k n}], x, x^{1/k}] \, \partial_x x^{1/k}$

Rule: If $n \in \mathbb{F}$, let $k \to \operatorname{Denominator}[n]$, then

$$\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx \longrightarrow k \operatorname{Subst}\left[\int x^{k-1} (a + b \operatorname{Log}[c (d + e x^{k n})^p])^q \, dx, x, x^{1/k}\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*Log[c*(d+e*x^(k*n))^p])^q,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e,p,q},x] && FractionQ[n]
```

U: $\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx$

Rule:

$$\int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx \longrightarrow \int (a + b \operatorname{Log}[c (d + e x^n)^p])^q \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_,x_Symbol] :=
  Unintegrable[(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,n,p,q},x]
```

N: $\int (a + b \operatorname{Log}[c\, v^p])^q \, dx$ when $v == d + e\, x^n$

Derivation: Algebraic normalization

Rule: If $v == d + e\, x^n$, then

$$\int (a + b \operatorname{Log}[c\, v^p])^q \, dx \longrightarrow \int (a + b \operatorname{Log}[c\, (d + e\, x^n)^p])^q \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*v_^p_.])^q_.,x_Symbol] :=
  Int[(a+b*Log[c*ExpandToSum[v,x]^p])^q,x] /;
FreeQ[{a,b,c,p,q},x] && BinomialQ[v,x] && Not[BinomialMatchQ[v,x]]
```

2. $\int (f\,x)^m \left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^q \,dx$

1. $\int (f\,x)^m \left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^q \,dx$ when $q == 1 \lor \left(\frac{m+1}{n} \in \mathbb{Z} \land \left(\frac{m+1}{n} > 0 \lor q \in \mathbb{Z}^+\right)\right)$

1: $\int x^m \left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^q \,dx$ when $\frac{m+1}{n} \in \mathbb{Z} \land \left(\frac{m+1}{n} > 0 \lor q \in \mathbb{Z}^+\right)$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n} \in \mathbb{Z}$, then $x^m\,F[x^n] == \frac{1}{n}\,\mathrm{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$

Rule: If $\frac{m+1}{n} \in \mathbb{Z} \land \left(\frac{m+1}{n} > 0 \lor q \in \mathbb{Z}^+\right)$, then

$$\int x^m \left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^q \,dx \to \frac{1}{n}\,\mathrm{Subst}\left[\int x^{\frac{m+1}{n}-1} \left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x\right)^p\right]\right)^q \,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Log[c*(d+e*x)^p])^q,x],x,x^n] /;
FreeQ[{a,b,c,d,e,m,n,p,q},x] && IntegerQ[Simplify[(m+1)/n]] && (GtQ[(m+1)/n,0] || IGtQ[q,0]) && Not[EqQ[q,1] && ILtQ[n,0] && IGtQ[m,0]]
```

2: $\int (f\,x)^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])\,dx$ when $m \neq -1$

Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'

Derivation: Integration by parts and piecewise constant extraction

Rule: If $m \neq -1$, then

$$\int (f\,x)^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])\,dx \longrightarrow \frac{(f\,x)^{m+1}\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])}{f\,(m+1)} - \frac{b\,e\,n\,p}{f\,(m+1)}\int\frac{x^{n-1}\,(f\,x)^{m+1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.]),x_Symbol] :=
  (f*x)^(m+1)*(a+b*Log[c*(d+e*x^n)^p])/(f*(m+1)) -
  b*e*n*p/(f*(m+1))*Int[x^(n-1)*(f*x)^(m+1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && NeQ[m,-1]
```

3: $\int (f\,x)^m \left(a+b \operatorname{Log}\left[c \left(d+e\,x^n\right)^p\right]\right)^q dx$ when $\frac{m+1}{n} \in \mathbb{Z} \land \left(\frac{m+1}{n} > 0 \lor q \in \mathbb{Z}^+\right)$

Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{(f\,x)^m}{x^m} == 0$

- Rule: If $\frac{m+1}{n} \in \mathbb{Z} \land \left(\frac{m+1}{n} > 0 \lor q \in \mathbb{Z}^+\right)$, then

$$\int (f\,x)^m \left(a+b \operatorname{Log}\left[c \left(d+e\,x^n\right)^p\right]\right)^q dx \to \frac{(f\,x)^m}{x^m} \int x^m \left(a+b \operatorname{Log}\left[c \left(d+e\,x^n\right)^p\right]\right)^q dx$$

Program code:

```
Int[(f_*x_)^m_*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  (f*x)^m/x^m*Int[x^m*(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && IntegerQ[Simplify[(m+1)/n]] && (GtQ[(m+1)/n,0] || IGtQ[q,0])
```

2: $\int (f\,x)^m \left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^q\,dx$ when $q - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z} \land m \neq -1$

Derivation: Integration by parts

Rule: If $q - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z} \land m \neq -1$, then

$$\int (f\,x)^m \left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^q\,dx \longrightarrow \frac{(f\,x)^{m+1}\left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^q}{f\,(m+1)} - \frac{b\,e\,n\,p\,q}{f^n\,(m+1)} \int \frac{(f\,x)^{m+n}\left(a + b\,\mathrm{Log}\left[c\,\left(d + e\,x^n\right)^p\right]\right)^{q-1}}{d + e\,x^n}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_,x_Symbol] :=
  (f*x)^(m+1)*(a+b*Log[c*(d+e*x^n)^p])^q/(f*(m+1)) -
  b*e*n*p*q/(f^n*(m+1))*Int[(f*x)^(m+n)*(a+b*Log[c*(d+e*x^n)^p])^(q-1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && IGtQ[q,1] && IntegerQ[n] && NeQ[m,-1]
```

3. $\int (f\,x)^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,dx$ when $n\in\mathbb{F}$

1: $\int x^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,dx$ when $n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $x^m\,F[x^n] == k\,\text{Subst}[x^{k\,(m+1)-1}\,F[x^{k\,n}],\,x,\,x^{1/k}]\,\partial_x x^{1/k}$

Rule: If $n\in\mathbb{F}$, let $k\to\text{Denominator}[n]$, then

$$\int x^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,dx \longrightarrow k\,\text{Subst}\Big[\int x^{k\,(m+1)-1}\,(a+b\,\text{Log}[c\,(d+e\,x^{k\,n})^p])^q\,dx,\,x,\,x^{1/k}\Big]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*Log[c*(d+e*x^(k*n))^p])^q,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e,m,p,q},x] && FractionQ[n]
```

2: $\int (f\,x)^m \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx$ when $n\in\mathbb{F}$

Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{(f\,x)^m}{x^m} == 0$

Rule: If $n\in\mathbb{F}$, then

$$\int (f\,x)^m \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx \longrightarrow \frac{(f\,x)^m}{x^m}\int x^m \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx$$

Program code:

```
Int[(f_*x_)^m_*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  (f*x)^m/x^m*Int[x^m*(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && FractionQ[n]
```

U: $\int (f\,x)^m \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx$

Rule:

$$\int (f\,x)^m \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx \longrightarrow \int (f\,x)^m \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x]
```

N: $\int (f\,x)^m\,(a+b\,\text{Log}[c\,v^p])^q\,dx$ when $v == d+e\,x^n$

Derivation: Algebraic normalization

Rule: If $v == d+e\,x^n$, then

$$\int (f\,x)^m\,(a+b\,\text{Log}[c\,v^p])^q\,dx \longrightarrow \int (f\,x)^m\,\left(a+b\,\text{Log}\left[c\,(d+e\,x^n)^p\right]\right)^q\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*Log[c_.*v_^p_.])^q_.,x_Symbol] :=
  Int[(f*x)^m*(a+b*Log[c*ExpandToSum[v,x]^p])^q,x] /;
FreeQ[{a,b,c,f,m,p,q},x] && BinomialQ[v,x] && Not[BinomialMatchQ[v,x]]
```

3. $\int (f+g\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,dx$

1. $\int (f+g\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])\,dx$ when $r\in\mathbb{Z}^+\lor n\in\mathbb{R}$

1: $\int \frac{a+b\,\text{Log}[c\,(d+e\,x^n)^p]}{f+g\,x}\,dx$ when $n\in\mathbb{R}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p]) == \frac{b\,e\,n\,p\,x^{n-1}}{d+e\,x^n}$

Rule: If $n\in\mathbb{R}$, then

$$\int \frac{a+b\,\text{Log}[c\,(d+e\,x^n)^p]}{f+g\,x}\,dx \to \frac{\text{Log}[f+g\,x]\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])}{g}-\frac{b\,e\,n\,p}{g}\int \frac{x^{n-1}\,\text{Log}[f+g\,x]}{d+e\,x^n}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])/(f_.+g_.*x_),x_Symbol] :=
  Log[f+g*x]*(a+b*Log[c*(d+e*x^n)^p])/g -
  b*e*n*p/g*Int[x^(n-1)*Log[f+g*x]/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && RationalQ[n]
```

2: $\int (f+g\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])\,dx$ when $(r\in\mathbb{Z}^+\lor n\in\mathbb{R})\land r\neq -1$

Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p]) == \frac{b\,e\,n\,p\,x^{n-1}}{d+e\,x^n}$

Rule: If $(r\in\mathbb{Z}^+\lor n\in\mathbb{R})\land r\neq -1$, then

$$\int (f+g\,x)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right) dx \longrightarrow \frac{(f+g\,x)^{r+1}\left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)}{g\,(r+1)} - \frac{b\,e\,n\,p}{g\,(r+1)} \int \frac{x^{n-1}\,(f+g\,x)^{r+1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.]),x_Symbol] :=
  (f+g*x)^(r+1)*(a+b*Log[c*(d+e*x^n)^p])/(g*(r+1)) -
  b*e*n*p/(g*(r+1))*Int[x^(n-1)*(f+g*x)^(r+1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,r},x] && (IGtQ[r,0] || RationalQ[n]) && NeQ[r,-1]
```

U: $\int (f+g\,x)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q dx$

Rule:

$$\int (f+g\,x)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q dx \longrightarrow \int (f+g\,x)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q dx$$

Program code:

```
Int[(f_.+g_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  Unintegrable[(f+g*x)^r*(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q,r},x]
```

N: $\int u^r\,(a+b\,\text{Log}[c\,v^p])^q\,dx$ when $u == f+g\,x\,\land\,v == d+e\,x^n$

Derivation: Algebraic normalization

Rule: If $u == f+g\,x\,\land\,v == d+e\,x^n$, then

$$\int u^r\,(a+b\,\text{Log}[c\,v^p])^q\,dx \longrightarrow \int (f+g\,x)^r\,\left(a+b\,\text{Log}[c\,(d+e\,x^n)^p]\right)^q\,dx$$

Program code:

```
Int[u_^r_.*(a_.+b_.*Log[c_.*v_^p_.])^q_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^r*(a+b*Log[c*ExpandToSum[v,x]^p])^q,x] /;
FreeQ[{a,b,c,p,q,r},x] && LinearQ[u,x] && BinomialQ[v,x] && Not[LinearMatchQ[u,x] && BinomialMatchQ[v,x]]
```

4. $\int (h\,x)^m\,(f+g\,x)^r\,\left(a+b\,\text{Log}[c\,(d+e\,x^n)^p]\right)^q\,dx$

1: $\int x^m\,(f+g\,x)^r\,\left(a+b\,\text{Log}[c\,(d+e\,x^n)^p]\right)^q\,dx$ when $m\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $m\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$, then

$$\int x^m\,(f+g\,x)^r\,\left(a+b\,\text{Log}[c\,(d+e\,x^n)^p]\right)^q\,dx \longrightarrow \int \left(a+b\,\text{Log}[c\,(d+e\,x^n)^p]\right)^q\,\text{ExpandIntegrand}[x^m\,(f+g\,x)^r,\,x]\,dx$$

Program code:

```
Int[x_^m_.*(f_.+g_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Log[c*(d+e*x^n)^p])^q,x^m*(f+g*x)^r,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q},x] && IntegerQ[m] && IntegerQ[r]
```

2: $\int (h x)^m (f+g x)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx$ when $m \in \mathbb{F} \land n \in \mathbb{Z} \land r \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $(h x)^m F[x] = \frac{k}{h} \operatorname{Subst}[x^{k(m+1)-1} F[\frac{x^k}{h}], x, (h x)^{1/k}] \partial_x (h x)^{1/k}$

Rule: If $m \in \mathbb{F} \land n \in \mathbb{Z} \land r \in \mathbb{Z}$, let $k = \operatorname{Denominator}[m]$, then

$$\int (h x)^m (f+g x)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx \to \frac{k}{h} \operatorname{Subst}\left[\int x^{k(m+1)-1} \left(f+\frac{g x^k}{h}\right)^r \left(a+b \operatorname{Log}\left[c \left(d+\frac{e x^{k n}}{h}\right)^p\right]\right)^q dx, x, (h x)^{1/k}\right]$$

Program code:

```
Int[(h_.*x_)^m_*(f_.+g_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_.)^p_.])^q_.,x_Symbol] :=
  With[{k=Denominator[m]},
  k/h*Subst[Int[x^(k*(m+1)-1)*(f+g*x^k/h)^r*(a+b*Log[c*(d+e*x^(k*n)/h^n)^p])^q,x],x,(h*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,h,p,r},x] && FractionQ[m] && IntegerQ[n] && IntegerQ[r]
```

U: $\int (h x)^m (f+g x)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx$

Rule:

$$\int (h x)^m (f+g x)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx \to \int (h x)^m (f+g x)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx$$

Program code:

```
Int[(h_.*x_)^m_.*(f_.+g_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  Unintegrable[(h*x)^m*(f+g*x)^r*(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q,r},x]
```

N: $\int (h\,x)^m u^r \left(a+b\,\mathrm{Log}\left[c\,v^p\right]\right)^q\,dx$ when $u == f+g\,x \wedge v == d+e\,x^n$

Derivation: Algebraic normalization

Rule: If $u == f+g\,x \wedge v == d+e\,x^n$, then

$$\int (h\,x)^m u^r \left(a+b\,\mathrm{Log}\left[c\,v^p\right]\right)^q\,dx \longrightarrow \int (h\,x)^m (f+g\,x)^r \left(a+b\,\mathrm{Log}\left[c\,(d+e\,x^n)^p\right]\right)^q\,dx$$

Program code:

```
Int[(h_.*x_)^m_.*u_^r_.*(a_.+b_.*Log[c_.*v_^p_.])^q_.,x_Symbol] :=
  Int[(h*x)^m*ExpandToSum[u,x]^r*(a+b*Log[c*ExpandToSum[v,x]^p])^q,x] /;
FreeQ[{a,b,c,h,m,p,q,r},x] && LinearQ[u,x] && BinomialQ[v,x] && Not[LinearMatchQ[u,x] && BinomialMatchQ[v,x]]
```

5. $\int (f+g\,x^s)^r \left(a+b\,\mathrm{Log}\left[c\,(d+e\,x^n)^p\right]\right)^q\,dx$

1: $\int \frac{a+b\,\mathrm{Log}\left[c\,(d+e\,x^n)^p\right]}{f+g\,x^2}\,dx$ when $n \in \mathbb{Z}$

Derivation: Integration by parts

Rule: If $n \in \mathbb{Z}$, let $u \to \int \frac{1}{f+g\,x^2}\,dx$, then

$$\int \frac{a+b\,\mathrm{Log}\left[c\,(d+e\,x^n)^p\right]}{f+g\,x^2}\,dx \longrightarrow u\left(a+b\,\mathrm{Log}\left[c\,(d+e\,x^n)^p\right]\right) - b\,e\,n\,p\int \frac{u\,x^{n-1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])/(f_+g_.*x_^2),x_Symbol] :=
  With[{u=IntHide[1/(f+g*x^2),x]},
  u*(a+b*Log[c*(d+e*x^n)^p]) - b*e*n*p*Int[u*x^(n-1)/(d+e*x^n),x]] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && IntegerQ[n]
```

2: $\int (f+g\,x^s)^r\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,dx$ when $n\in\mathbb{Z}\wedge q\in\mathbb{Z}^+\wedge r\in\mathbb{Z}\wedge s\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}\wedge q\in\mathbb{Z}^+\wedge r\in\mathbb{Z}\wedge s-1\in\mathbb{Z}^+$, then

$$\int (f+g\,x^s)^r\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,dx\;\to\;\int (a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,\text{ExpandIntegrand}[(f+g\,x^s)^r,\,x]\,dx$$

Program code:

```
Int[(f_+g_.*x_^s_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  With[{t=ExpandIntegrand[(a+b*Log[c*(d+e*x^n)^p])^q,(f+g*x^s)^r,x]},
  Int[t,x] /;
 SumQ[t]] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q,r,s},x] && IntegerQ[n] && IGtQ[q,0] && IntegerQ[r] && IntegerQ[s] &&
  (EqQ[q,1] || GtQ[r,0] && GtQ[s,1] || LtQ[s,0] && LtQ[r,0])
```

3: $\int (f+g\,x^s)^r\,(a+b\,\mathrm{Log}[c\,(d+e\,x^n)^p])^q\,dx$ when $n\in\mathbb{F}\wedge s\,\mathrm{Denominator}[n]\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $F[x^n]=k\,\mathrm{Subst}[x^{k-1}\,F[x^{k\,n}],\,x,\,x^{1/k}]\,\partial_x x^{1/k}$

Rule: If $n\in\mathbb{F}$, let $k\to\mathrm{Denominator}[n]$, if $k\,s\in\mathbb{Z}$, then

$$\int (f+g\,x^s)^r\,(a+b\,\mathrm{Log}[c\,(d+e\,x^n)^p])^q\,dx\;\longrightarrow\;k\,\mathrm{Subst}\Big[\int x^{k-1}\,(f+g\,x^{k\,s})^r\,(a+b\,\mathrm{Log}[c\,(d+e\,x^{k\,n})^p])^q\,dx,\,x,\,x^{1/k}\Big]$$

Program code:

```
Int[(f_+g_.*x_^s_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(f+g*x^(k*s))^r*(a+b*Log[c*(d+e*x^(k*n))^p])^q,x],x,x^(1/k)] /;
 IntegerQ[k*s]] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q,r,s},x] && FractionQ[n]
```

U: $\int (f+g\,x^s)^r\,(a+b\,\mathrm{Log}[c\,(d+e\,x^n)^p])^q\,dx$

Rule:

$$\int (f+g\,x^s)^r\,(a+b\,\mathrm{Log}[c\,(d+e\,x^n)^p])^q\,dx\;\longrightarrow\;\int (f+g\,x^s)^r\,(a+b\,\mathrm{Log}[c\,(d+e\,x^n)^p])^q\,dx$$

Program code:

```
Int[(f_+g_.*x_^s_)^r_.(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  Unintegrable[(f+g*x^s)^r*(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q,r,s},x]
```

N: $\int u^r\,(a+b\,\text{Log}[c\,v^p])^q\,dx$ when $u == f+g\,x^s\ \land\ v == d+e\,x^n$

Derivation: Algebraic normalization

Rule: If $u == f+g\,x^s\ \land\ v == d+e\,x^n$, then

$$\int u^r\,(a+b\,\text{Log}[c\,v^p])^q\,dx \longrightarrow \int (f+g\,x^s)^r\,\left(a+b\,\text{Log}\left[c\,(d+e\,x^n)^p\right]\right)^q\,dx$$

Program code:

```
Int[u_^r_.*(a_.+b_.*Log[c_.*v_^p_.])^q_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^r*(a+b*Log[c*ExpandToSum[v,x]^p])^q,x] /;
FreeQ[{a,b,c,p,q,r},x] && BinomialQ[{u,v},x] && Not[BinomialMatchQ[{u,v},x]]
```

6. $\int (h\,x)^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q \,dx$

1: $\int x^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q \,dx$ when $r\in\mathbb{Z}\,\land\,\frac{s}{n}\in\mathbb{Z}\,\land\,\frac{m+1}{n}\in\mathbb{Z}\,\land\,\left(\frac{m+1}{n}>0\,\lor\,q\in\mathbb{Z}^+\right)$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F[x^n] == \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x\,x^n$

Rule: If $r\in\mathbb{Z}\,\land\,\frac{s}{n}\in\mathbb{Z}\,\land\,\frac{m+1}{n}\in\mathbb{Z}\,\land\,\left(\frac{m+1}{n}>0\,\lor\,q\in\mathbb{Z}^+\right)$, then

$$\int x^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q \,dx \,\longrightarrow\, \frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\left(f+g\,x^{\frac{s}{n}}\right)^r \left(a+b\,\text{Log}\left[c\,\left(d+e\,x\right)^p\right]\right)^q \,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(f_+g_.*x_^s_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(f+g*x^(s/n))^r*(a+b*Log[c*(d+e*x)^p])^q,x],x,x^n] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r,s},x] && IntegerQ[r] && IntegerQ[s/n] && IntegerQ[Simplify[(m+1)/n]] && (GtQ[(m+1)/n,0] || IGtQ[q,0])
```

2: $\int x^m (f+g x^s)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx$ when $q \in \mathbb{Z}^+ \land m \in \mathbb{Z} \land r \in \mathbb{Z} \land s \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $q \in \mathbb{Z}^+ \land m \in \mathbb{Z} \land r \in \mathbb{Z} \land s \in \mathbb{Z}$, then

$$\int x^m (f+g x^s)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx \to \int (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, \text{ExpandIntegrand}[x^m (f+g x^s)^r, x] \, dx$$

Program code:

```
Int[x_^m_.*(f_+g_.*x_^s_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Log[c*(d+e*x^n)^p])^q,x^m*(f+g*x^s)^r,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r,s},x] && IGtQ[q,0] && IntegerQ[m] && IntegerQ[r] && IntegerQ[s]
```

?: $\int x^m (f+g x^s)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx$ when $n \in \mathbb{F} \land m \, \text{Denominator}[n] \in \mathbb{Z} \land s \, \text{Denominator}[n] \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x^n] = k \, \text{Subst}[x^{k-1} F[x^{k\,n}], x, x^{1/k}] \, \partial_x x^{1/k}$

Rule: If $n \in \mathbb{F}$, let $k \to \text{Denominator}[n]$, if $k\,m \in \mathbb{Z} \land k\,s \in \mathbb{Z}$, then

$$\int x^m (f+g x^s)^r (a+b \operatorname{Log}[c (d+e x^n)^p])^q \, dx \to k \, \text{Subst}\left[\int x^{k-1} (f+g x^{k\,s})^r (a+b \operatorname{Log}[c (d+e x^{k\,n})^p])^q \, dx, x, x^{1/k}\right]$$

Program code:

```
Int[(f_+g_.*x_^s_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(f+g*x^(k*s))^r*(a+b*Log[c*(d+e*x^(k*n))^p])^q,x],x,x^(1/k)] /;
 IntegerQ[k*s]] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q,r,s},x] && FractionQ[n]
```

3: $\int x^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx$ when $n\in\mathbb{F}\land\frac{1}{n}\in\mathbb{Z}\land\frac{s}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{1}{n}\in\mathbb{Z}$, then $F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{1}{n}-1}\,F[x],\,x,\,x^n\right]\partial_x x^n$

Rule: If $n\in\mathbb{F}\land\frac{1}{n}\in\mathbb{Z}\land\frac{s}{n}\in\mathbb{Z}$, then

$$\int x^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\,\left(d+e\,x^n\right)^p\right]\right)^q\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int x^{m+\frac{1}{n}-1} \left(f+g\,x^{s/n}\right)^r \left(a+b\,\text{Log}\left[c\,\left(d+e\,x\right)^p\right]\right)^q\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(f_+g_.*x_^s_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  1/n*Subst[Int[x^(m+1/n-1)*(f+g*x^(s/n))^r*(a+b*Log[c*(d+e*x)^p])^q,x],x,x^n] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q,r,s},x] && FractionQ[n] && IntegerQ[1/n] && IntegerQ[s/n]
```

4: $\int (h\,x)^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q\,dx$ when $m\in\mathbb{F}\land n\in\mathbb{Z}\land s\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $(h\,x)^m\,F[x]==\frac{k}{h}\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{h}\right],\,x,\,(h\,x)^{1/k}\right]\,\partial_x\,(h\,x)^{1/k}$

Rule: If $m\in\mathbb{F}\land n\in\mathbb{Z}\land s\in\mathbb{Z}$, let $k=\text{Denominator}[m]$, then

$$\int (h\,x)^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q\,dx \rightarrow \frac{k}{h}\,\text{Subst}\left[\int x^{k\,(m+1)-1}\left(f+\frac{g\,x^{k\,s}}{h}\right)^r\left(a+b\,\text{Log}\left[c\left(d+\frac{e\,x^{k\,n}}{h}\right)^p\right]\right)^q dx,\,x,\,(h\,x)^{1/k}\right]$$

Program code:

```
Int[(h_.*x_)^m_*(f_.+g_.*x_^s_.)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_.)^p_.])^q_.,x_Symbol] :=
  With[{k=Denominator[m]},
  k/h*Subst[Int[x^(k*(m+1)-1)*(f+g*x^(k*s)/h^s)^r*(a+b*Log[c*(d+e*x^(k*n)/h^n)^p])^q,x],x,(h*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,h,p,r},x] && FractionQ[m] && IntegerQ[n] && IntegerQ[s]
```

U: $\int (h\,x)^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q\,dx$

Rule:

$$\int (h\,x)^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q\,dx \rightarrow \int (h\,x)^m \left(f+g\,x^s\right)^r \left(a+b\,\text{Log}\left[c\left(d+e\,x^n\right)^p\right]\right)^q\,dx$$

Program code:

```
Int[(h_.*x_)^m_.*(f_+g_.*x_^s_)^r_.(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])^q_.,x_Symbol] :=
  Unintegrable[(h*x)^m*(f+g*x^s)^r*(a+b*Log[c*(d+e*x^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q,r,s},x]
```

N: $\int (h\,x)^m\,u^r\,(a+b\,\text{Log}[c\,v^p])^q\,dx$ when $u == f+g\,x^s \wedge v == d+e\,x^n$

Derivation: Algebraic normalization

Rule: If $u == f+g\,x^s \wedge v == d+e\,x^n$, then

$$\int (h\,x)^m\,u^r\,(a+b\,\text{Log}[c\,v^p])^q\,dx \;\longrightarrow\; \int (h\,x)^m\,(f+g\,x^s)^r\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])^q\,dx$$

Program code:

```
Int[(h_.*x_)^m_.*u_^r_.*(a_.+b_.*Log[c_.*v_^p_.])^q_.,x_Symbol] :=
  Int[(h*x)^m*ExpandToSum[u,x]^r*(a+b*Log[c*ExpandToSum[v,x]^p])^q,x] /;
FreeQ[{a,b,c,h,m,p,q,r},x] && BinomialQ[{u,v},x] && Not[BinomialMatchQ[{u,v},x]]
```

7: $\int \frac{\text{Log}[f\,x^q]^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])}{x}\,dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\frac{\text{Log}[c\,x^q]^m}{x} == \partial_x \frac{\text{Log}[c\,x^q]^{m+1}}{q\,(m+1)}$

Rule: If $m \neq -1$, then

$$\int \frac{\text{Log}[f\,x^q]^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])}{x}\,dx \longrightarrow \frac{\text{Log}[f\,x^q]^{m+1}\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])}{q\,(m+1)} - \frac{b\,e\,n\,p}{q\,(m+1)} \int \frac{x^{n-1}\,\text{Log}[f\,x^q]^{m+1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[Log[f_.*x_^q_.]^m_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.])/x_,x_Symbol] :=
  Log[f*x^q]^(m+1)*(a+b*Log[c*(d+e*x^n)^p])/(q*(m+1)) -
  b*e*n*p/(q*(m+1))*Int[x^(n-1)*Log[f*x^q]^(m+1)/(d+e*x^n),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q},x] && NeQ[m,-1]
```

8: $\int \text{ArcTrig}[f\,x]^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])\,dx$ when $m\in\mathbb{Z}^+\land n-1\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $m\in\mathbb{Z}^+\land n-1\in\mathbb{Z}^+$, let $u\to\int \text{ArcTrig}[f\,x]^m\,dx$, then

$$\int \text{ArcTrig}[f\,x]^m\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])\,dx\;\longrightarrow\;u\,(a+b\,\text{Log}[c\,(d+e\,x^n)^p])-b\,e\,n\,p\int\frac{u\,x^{n-1}}{d+e\,x^n}\,dx$$

Program code:

```
Int[F_[f_.*x_]^m_.*(a_.+b_.*Log[c_.*(d_+e_.*x_^n_)^p_.]),x_Symbol] :=
  With[{u=IntHide[F[f*x]^m,x]},
  Dist[a+b*Log[c*(d+e*x^n)^p],u,x] - b*e*n*p*Int[SimplifyIntegrand[u*x^(n-1)/(d+e*x^n),x],x]] /;
FreeQ[{a,b,c,d,e,f,p},x] && MemberQ[{ArcSin,ArcCos,ArcSinh,ArcCosh},F] && IGtQ[m,0] && IGtQ[n,1]
```

Rules for integrands of the form $u\,(a+b\,\mathrm{Log}[c\,(d+e\,x^n)^p])^q$

1: $\int (a+b\,\mathrm{Log}[c\,(d+e\,(f+g\,x)^n)^p])^q\,dx$ when $q\in\mathbb{Z}^+\land(q==1\lor n\in\mathbb{Z})$

Derivation: Integration by substitution

Rule: If $q\in\mathbb{Z}^+\land(q==1\lor n\in\mathbb{Z})$, then

$$\int (a+b\,\mathrm{Log}[c\,(d+e\,(f+g\,x)^n)^p])^q\,dx \to \frac{1}{g}\,\mathrm{Subst}\Big[\int (a+b\,\mathrm{Log}[c\,(d+e\,x^n)^p])^q\,dx,\ x,\ f+g\,x\Big]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*(f_.+g_.*x_)^n_)^p_.])^q_.,x_Symbol] :=
  1/g*Subst[Int[(a+b*Log[c*(d+e*x^n)^p])^q,x],x,f+g*x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && IGtQ[q,0] && (EqQ[q,1] || IntegerQ[n])
```

U: $\int \left(a+b \operatorname{Log}\left[c \left(d+e \left(f+g x\right)^n\right)^p\right]\right)^q \, dx$

Rule:

$$\int \left(a+b \operatorname{Log}\left[c \left(d+e \left(f+g x\right)^n\right)^p\right]\right)^q \, dx \longrightarrow \int \left(a+b \operatorname{Log}\left[c \left(d+e \left(f+g x\right)^n\right)^p\right]\right)^q \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*(f_.+g_.*x_)^n_)^p_.])^q_.,x_Symbol] :=
  Unintegrable[(a+b*Log[c*(d+e*(f+g*x)^n)^p])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q},x]
```