Rules for integrands of the form $(d\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n$

1. $\int (d\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

1: $\int \frac{(a+b\,\text{ArcSin}[c\,x])^n}{x}\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $\frac{F[\text{ArcSin}[c\,x]]}{x} == \text{Subst}\left[\frac{F[x]}{\text{Tan}[x]},\,x,\,\text{ArcSin}[c\,x]\right]\,\partial_x \text{ArcSin}[c\,x]$

Note: $\frac{(a+b\,x)^n}{\text{Tan}[x]}$ is not integrable unless $n\in\mathbb{Z}^+$.

Rule: If $n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{ArcSin}[c\,x])^n}{x}\,dx \to \text{Subst}\left[\int \frac{(a+b\,x)^n}{\text{Tan}[x]}\,dx,\,x,\,\text{ArcSin}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_./x_,x_Symbol] :=
  Subst[Int[(a+b*x)^n/Tan[x],x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_./x_,x_Symbol] :=
  -Subst[Int[(a+b*x)^n/Cot[x],x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[n,0]
```

2: $\int (d\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\land m\neq -1$

Reference: G&R 2.831, CRC 453, A&S 4.4.65

Reference: G&R 2.832, CRC 454, A&S 4.4.67

Derivation: Integration by parts

Basis: $\partial_x (a+b\,\text{ArcSin}[c\,x])^n = \frac{b\,c\,n\,(a+b\,\text{ArcSin}[c\,x])^{n-1}}{\sqrt{1-c^2 x^2}}$

Rule: If $n \in \mathbb{Z}^+ \land m \neq -1$, then

$$\int (d\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx \to \frac{(d\,x)^{m+1}\,(a+b\,\text{ArcSin}[c\,x])^n}{d\,(m+1)} - \frac{b\,c\,n}{d\,(m+1)}\int \frac{(d\,x)^{m+1}\,(a+b\,\text{ArcSin}[c\,x])^{n-1}}{\sqrt{1-c^2 x^2}}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcSin[c*x])^n/(d*(m+1)) -
  b*c*n/(d*(m+1))*Int[(d*x)^(m+1)*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcCos[c*x])^n/(d*(m+1)) +
  b*c*n/(d*(m+1))*Int[(d*x)^(m+1)*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

2. $\int x^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $m \in \mathbb{Z}^+$

1: $\int x^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $m \in \mathbb{Z}^+ \land n > 0$

Reference: G&R 2.831, CRC 453, A&S 4.4.65

Reference: G&R 2.832, CRC 454, A&S 4.4.67

Derivation: Integration by parts

Basis: $\partial_x (a+b\,\text{ArcSin}[c\,x])^n = \frac{b\,c\,n\,(a+b\,\text{ArcSin}[c\,x])^{n-1}}{\sqrt{1-c^2 x^2}}$

Rule: If $n \in \mathbb{Z}^+ \land m \neq -1$, then

$$\int x^m \left(a + b\,\text{ArcSin}[c\,x]\right)^n dx \longrightarrow \frac{x^{m+1}\left(a + b\,\text{ArcSin}[c\,x]\right)^n}{m+1} - \frac{b\,c\,n}{m+1}\int \frac{x^{m+1}\left(a + b\,\text{ArcSin}[c\,x]\right)^{n-1}}{\sqrt{1 - c^2\,x^2}}\,dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  x^(m+1)*(a+b*ArcSin[c*x])^n/(m+1) -
  b*c*n/(m+1)*Int[x^(m+1)*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GtQ[n,0]
```

```
Int[x_^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  x^(m+1)*(a+b*ArcCos[c*x])^n/(m+1) +
  b*c*n/(m+1)*Int[x^(m+1)*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GtQ[n,0]
```

2. $\int x^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\land n<-1$

1: $\int x^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\land -2\le n<-1$

Derivation: Integration by parts and integration by substitution

Basis: $\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{1-c^2x^2}} = \partial_x\frac{(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: $\frac{F[x]}{\sqrt{1-c^2x^2}} = \frac{1}{c}\,\text{Subst}\left[F\left[\frac{\text{Sin}[x]}{c}\right],x,\text{ArcSin}[c\,x]\right]\partial_x\text{ArcSin}[c\,x]$

Basis: If $c>0\lor m\in\mathbb{Z}$, then
$\frac{x^{m-1}\,(m-(m+1)\,c^2x^2)}{\sqrt{1-c^2x^2}} = \frac{1}{c^m}\,\text{Subst}\left[\text{Sin}[x]^{m-1}\left(m-(m+1)\,\text{Sin}[x]^2\right),x,\text{ArcSin}[c\,x]\right]\partial_x\text{ArcSin}[c\,x]$

Note: Although not essential, by switching to the trig world this rule saves numerous steps and results in more compact antiderivatives.

Rule: If $m\in\mathbb{Z}^+\land -2\le n<-1$, then

$$\int x^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx\;\longrightarrow$$

$$\frac{x^m\sqrt{1-c^2x^2}\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)}-\frac{1}{b\,c\,(n+1)}\int\frac{x^{m-1}\left(m-(m+1)\,c^2x^2\right)(a+b\,\text{ArcSin}[c\,x])^{n+1}}{\sqrt{1-c^2x^2}}\,dx\;\longrightarrow$$

$$\frac{x^m\sqrt{1-c^2x^2}\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)}-\frac{1}{b\,c^{m+1}\,(n+1)}\,\text{Subst}\left[\int(a+b\,x)^{n+1}\,\text{Sin}[x]^{m-1}\left(m-(m+1)\,\text{Sin}[x]^2\right)dx,x,\text{ArcSin}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1-c^2*x^2]*(a+b*ArcSin[c*x])^(n+1)/(b*c*(n+1)) -
  1/(b*c^(m+1)*(n+1))*Subst[Int[ExpandTrigReduce[(a+b*x)^(n+1),Sin[x]^(m-1)*(m-(m+1)*Sin[x]^2),x],x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GeQ[n,-2] && LtQ[n,-1]
```

```
Int[x_^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -x^m*Sqrt[1-c^2*x^2]*(a+b*ArcCos[c*x])^(n+1)/(b*c*(n+1)) -
  1/(b*c^(m+1)*(n+1))*Subst[Int[ExpandTrigReduce[(a+b*x)^(n+1),Cos[x]^(m-1)*(m-(m+1)*Cos[x]^2),x],x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GeQ[n,-2] && LtQ[n,-1]
```

2: $\int x^m (a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\wedge n<-2$

Derivation: Integration by parts and algebraic expansion

Basis: $\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{1-c^2 x^2}} == \partial_x \frac{(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: $\partial_x \left(x^m \sqrt{1-c^2 x^2}\right) == \frac{m\,x^{m-1}}{\sqrt{1-c^2 x^2}} - \frac{c^2 (m+1)\,x^{m+1}}{\sqrt{1-c^2 x^2}}$

Rule: If $m\in\mathbb{Z}^+\wedge n<-2$, then

$$\int x^m (a+b\,\text{ArcSin}[c\,x])^n\,dx \to$$
$$\frac{x^m \sqrt{1-c^2 x^2}\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)} - \frac{m}{b\,c\,(n+1)}\int \frac{x^{m-1}(a+b\,\text{ArcSin}[c\,x])^{n+1}}{\sqrt{1-c^2 x^2}}\,dx + \frac{c\,(m+1)}{b\,(n+1)}\int \frac{x^{m+1}(a+b\,\text{ArcSin}[c\,x])^{n+1}}{\sqrt{1-c^2 x^2}}\,dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1-c^2*x^2]*(a+b*ArcSin[c*x])^(n+1)/(b*c*(n+1)) -
  m/(b*c*(n+1))*Int[x^(m-1)*(a+b*ArcSin[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] +
  c*(m+1)/(b*(n+1))*Int[x^(m+1)*(a+b*ArcSin[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && LtQ[n,-2]
```

```
Int[x_^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -x^m*Sqrt[1-c^2*x^2]*(a+b*ArcCos[c*x])^(n+1)/(b*c*(n+1)) +
  m/(b*c*(n+1))*Int[x^(m-1)*(a+b*ArcCos[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] -
  c*(m+1)/(b*(n+1))*Int[x^(m+1)*(a+b*ArcCos[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && LtQ[n,-2]
```

3: $\int x^m (a + b\,\text{ArcSin}[c\,x])^n\,dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}^+$, then $x^m\,F[\text{ArcSin}[c\,x]] = \frac{1}{c^{m+1}}\,\text{Subst}[F[x]\,\text{Sin}[x]^m\,\text{Cos}[x],\,x,\,\text{ArcSin}[c\,x]]\,\partial_x\text{ArcSin}[c\,x]$

Note: If $m \in \mathbb{Z}^+$, then $(a + b\,x)^n\,\text{Sin}[x]^m\,\text{Cos}[x]$ is integrable in closed-form.

Rule: If $m \in \mathbb{Z}^+$, then

$$\int x^m (a + b\,\text{ArcSin}[c\,x])^n\,dx \longrightarrow \frac{1}{c^{m+1}}\,\text{Subst}\left[\int (a + b\,x)^n\,\text{Sin}[x]^m\,\text{Cos}[x]\,dx,\,x,\,\text{ArcSin}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Sin[x]^m*Cos[x],x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[m,0]
```

```
Int[x_^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -1/c^(m+1)*Subst[Int[(a+b*x)^n*Cos[x]^m*Sin[x],x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[m,0]
```

U: $\int (d\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$

Rule:

$$\int (d\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx \longrightarrow \int (d\,x)^m\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```