Rules for integrands of the form $P[x]\ (a+b\,x)^m\ (c+d\,x)^n\ (e+f\,x)^p\ (g+h\,x)^q$

1. $\int \frac{(a+b\,x)^m\ (A+B\,x+C\,x^2)}{\sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx$ when $2\,m\in\mathbb{Z}\ \land\ m>-1$

   1: $\int \frac{A+B\,x+C\,x^2}{\sqrt{a+b\,x}\ \sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx$

Rule:

$$\int \frac{A+B\,x+C\,x^2}{\sqrt{a+b\,x}\ \sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx\ \longrightarrow$$

$$\frac{C\,\sqrt{a+b\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}{b\,f\,h\,\sqrt{c+d\,x}}+$$

$$\frac{C\,(d\,e-c\,f)\,(d\,g-c\,h)}{2\,b\,d\,f\,h}\int \frac{\sqrt{a+b\,x}}{(c+d\,x)^{3/2}\,\sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx+$$

$$\frac{1}{2\,b\,d\,f\,h}\int \Big(\big(2\,A\,b\,d\,f\,h-C\,(b\,d\,e\,g+a\,c\,f\,h)+(2\,b\,B\,d\,f\,h-C\,(a\,d\,f\,h+b\,(d\,f\,g+d\,e\,h+c\,f\,h)))\,x\big)\Big/\left(\sqrt{a+b\,x}\ \sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}\right)\Big)\,dx$$

Program code:

```
Int[(A_.+B_.*x_+C_.*x_^2)/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  C*Sqrt[a+b*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(b*f*h*Sqrt[c+d*x]) +
  C*(d*e-c*f)*(d*g-c*h)/(2*b*d*f*h)*Int[Sqrt[a+b*x]/((c+d*x)^(3/2)*Sqrt[e+f*x]*Sqrt[g+h*x]),x] +
  1/(2*b*d*f*h)*Int[1/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x])*
     Simp[2*A*b*d*f*h-C*(b*d*e*g+a*c*f*h)+(2*b*B*d*f*h-C*(a*d*f*h+b*(d*f*g+d*e*h+c*f*h)))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,A,B,C},x]
```

```
Int[(A_.+C_.*x_^2)/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  C*Sqrt[a+b*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(b*f*h*Sqrt[c+d*x]) +
  C*(d*e-c*f)*(d*g-c*h)/(2*b*d*f*h)*Int[Sqrt[a+b*x]/((c+d*x)^(3/2)*Sqrt[e+f*x]*Sqrt[g+h*x]),x] +
  1/(2*b*d*f*h)*Int[1/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x])*
     Simp[2*A*b*d*f*h-C*(b*d*e*g+a*c*f*h)-C*(a*d*f*h+b*(d*f*g+d*e*h+c*f*h))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,A,C},x]
```

2: $\int \frac{(a+b\,x)^m\,\left(A+B\,x+C\,x^2\right)}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$ when $2\,m\in\mathbb{Z}\,\wedge\,m>0$

Rule: If $2\,m\in\mathbb{Z}\,\wedge\,m>0$, then

$$\int \frac{(a+b\,x)^m\,\left(A+B\,x+C\,x^2\right)}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx\,\longrightarrow$$

$$\frac{2\,C\,(a+b\,x)^m\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{d\,f\,h\,(2\,m+3)}+$$
$$\frac{1}{d\,f\,h\,(2\,m+3)}\int\frac{(a+b\,x)^{m-1}}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\cdot$$
$$\big(a\,A\,d\,f\,h\,(2\,m+3)-C\,\left(a\,\left(d\,e\,g+c\,f\,g+c\,e\,h\right)+2\,b\,c\,e\,g\,m\right)+$$
$$\left(\left(A\,b+a\,B\right)\,d\,f\,h\,(2\,m+3)-C\,\left(2\,a\,\left(d\,f\,g+d\,e\,h+c\,f\,h\right)+b\,(2\,m+1)\,\left(d\,e\,g+c\,f\,g+c\,e\,h\right)\right)\right)\,x+$$
$$\left(b\,B\,d\,f\,h\,(2\,m+3)+2\,C\,\left(a\,d\,f\,h\,m-b\,(m+1)\,\left(d\,f\,g+d\,e\,h+c\,f\,h\right)\right)\right)\,x^2\big)\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_./(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2]},
  2*C*(a+b*x)^m*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(d*f*h*(2*m+3)) +
  1/(d*f*h*(2*m+3))*Int[((a+b*x)^(m-1)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
     Simp[a*A*d*f*h*(2*m+3)-C*(a*(d*e*g+c*f*g+c*e*h)+2*b*c*e*g*m) +
      ((A*b+a*B)*d*f*h*(2*m+3)-C*(2*a*(d*f*g+d*e*h+c*f*h)+b*(2*m+1)*(d*e*g+c*f*g+c*e*h)))*x +
      (b*B*d*f*h*(2*m+3)+2*C*(a*d*f*h*m-b*(m+1)*(d*f*g+d*e*h+c*f*h)))*x^2,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && PolyQ[Px,x] && LeQ[1,Expon[Px,x],2] && IntegerQ[2*m] && GtQ[m,0]
```

2: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$ when $\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x] == 0$

Derivation: Algebraic expansion

Basis: If $\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x] == 0$, then
$P[x] == (a+b\,x)\,\text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]$

Rule: If $\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x] == 0$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx \to \int \text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]\,(a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.*(g_.+h_.*x_)^q_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,a+b*x,x]*(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+b*x,x],0] && EqQ[m,-1]
```

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.*(g_.+h_.*x_)^q_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,a+b*x,x]*(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+b*x,x],0]
```

3: $\int \frac{(a+b\,x)^m\,(A+B\,x+C\,x^2)}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$ when $2\,m \in \mathbb{Z}\,\wedge\,m<-1$

Rule: If $2\,m \in \mathbb{Z}\,\wedge\,m<-1$, then

$$\int \frac{(a+b\,x)^m\,(A+B\,x+C\,x^2)}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx \to$$

$$\frac{\left(A b^2 - a b B + a^2 C\right) (a+b x)^{m+1} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}}{(m+1) (b c-a d) (b e-a f) (b g-a h)} - \frac{1}{2 (m+1) (b c-a d) (b e-a f) (b g-a h)} \int \frac{(a+b x)^{m+1}}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \cdot$$
$$\Big(A \left(2 a^2 d f h (m+1) - 2 a b (m+1) (d f g+d e h+c f h) + b^2 (2 m+3) (d e g+c f g+c e h)\right) - (b B-a C) \left(a (d e g+c f g+c e h) + 2 b c e g (m+1)\right) -$$
$$2 \left((A b-a B) \left(a d f h (m+1) - b (m+2) (d f g+d e h+c f h)\right) - C \left(a^2 (d f g+d e h+c f h) - b^2 c e g (m+1) + a b (m+1) (d e g+c f g+c e h)\right)\right) x +$$
$$d f h (2 m+5) \left(A b^2 - a b B + a^2 C\right) x^2\Big)\, dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_/(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,1],C=Coeff[Px,x,2]},
  (A*b^2-a*b*B+a^2*C)*(a+b*x)^(m+1)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/((m+1)*(b*c-a*d)*(b*e-a*f)*(b*g-a*h)) -
  1/(2*(m+1)*(b*c-a*d)*(b*e-a*f)*(b*g-a*h))*Int[((a+b*x)^(m+1)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
    Simp[A*(2*a^2*d*f*h*(m+1)-2*a*b*(m+1)*(d*f*g+d*e*h+c*f*h)+b^2*(2*m+3)*(d*e*g+c*f*g+c*e*h)) -
       (b*B-a*C)*(a*(d*e*g+c*f*g+c*e*h)+2*b*c*e*g*(m+1)) -
       2*((A*b-a*B)*(a*d*f*h*(m+1)-b*(m+2)*(d*f*g+d*e*h+c*f*h))-C*(a^2*(d*f*g+d*e*h+c*f*h)-b^2*c*e*g*(m+1)+a*b*(m+1)*(d*e*g+c*f*
      d*f*h*(2*m+5)*(A*b^2-a*b*B+a^2*C)*x^2,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && PolyQ[Px,x] && LeQ[1,Expon[Px,x],2] && IntegerQ[2*m] && LtQ[m,-1]
```

4: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$ when $(m\mid n)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx\ \longrightarrow\ \int \text{ExpandIntegrand}\left[P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q,\ x\right]dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.*(g_.+h_.*x_)^q_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x] && PolyQ[Px,x] && IntegersQ[m,n]
```

5: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$

Derivation: Algebraic expansion

Basis:
$P[x]==$
$\text{PolynomialRemainder}[P[x],\ a+b\,x,\ x]+(a+b\,x)\,\text{PolynomialQuotient}[P[x],\ a+b\,x,\ x]$

Note: Reduces the degree of the polynomial, but results in exponential growth.

Rule:

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx\ \longrightarrow$$

$$\text{PolynomialRemainder}[P[x],\ a+b\,x,\ x]\int(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx\ +$$

$$\int \text{PolynomialQuotient}[P[x],\, a+b\,x,\, x]\,(a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.*(g_.+h_.*x_)^q_.,x_Symbol] :=
  PolynomialRemainder[Px,a+b*x,x]*Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x] +
  Int[PolynomialQuotient[Px,a+b*x,x]*(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x] && PolyQ[Px,x] && EqQ[m,-1]
```

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.*(g_.+h_.*x_)^q_.,x_Symbol] :=
  PolynomialRemainder[Px,a+b*x,x]*Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x] +
  Int[PolynomialQuotient[Px,a+b*x,x]*(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x] && PolyQ[Px,x]
```

Rules for integrands of the form $P[x]\ (a+b\,x)^m\ (c+d\,x)^n\ (e+f\,x)^p$

1: $\displaystyle\int\frac{A+B\,x+C\,x^2}{\sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx$

Rule:

$$\int\frac{A+B\,x+C\,x^2}{\sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx\ \longrightarrow$$

$$\frac{2\,C\,\sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}{3\,d\,f\,h}+$$

$$\frac{1}{3\,d\,f\,h}\int\Big(\big(3\,A\,d\,f\,h-C\,(d\,e\,g+c\,f\,g+c\,e\,h)+(3\,B\,d\,f\,h-2\,C\,(d\,f\,g+d\,e\,h+c\,f\,h))\,x\big)\Big/\Big(\sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}\Big)\Big)\,dx$$

Program code:

```
Int[(A_.+B_.*x_+C_.*x_^2)/(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  2*C*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(3*d*f*h) +
  1/(3*d*f*h)*Int[1/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x])*
     Simp[3*A*d*f*h-C*(d*e*g+c*f*g+c*e*h)+(3*B*d*f*h-2*C*(d*f*g+d*e*h+c*f*h))*x,x],x] /;
FreeQ[{c,d,e,f,g,h,A,B,C},x]
```

```
Int[(A_.+C_.*x_^2)/(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  2*C*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(3*d*f*h) +
  1/(3*d*f*h)*Int[1/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x])*
     Simp[3*A*d*f*h-C*(d*e*g+c*f*g+c*e*h)-2*C*(d*f*g+d*e*h+c*f*h)*x,x],x] /;
FreeQ[{c,d,e,f,g,h,A,C},x]
```

2. $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $b\,c+a\,d=0\land m=n$

1: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $b\,c+a\,d=0\land m=n\land(m\in\mathbb{Z}\lor a>0\land c>0)$

Derivation: Algebraic simplification

Basis: If $b\,c+a\,d=0\land(m\in\mathbb{Z}\lor a>0\land c>0)$, then $(a+b\,x)^m\,(c+d\,x)^m=(a\,c+b\,d\,x^2)^m$

Rule: If $b\,c+a\,d=0\land m=n\land(m\in\mathbb{Z}\lor a>0\land c>0)$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow\;\int P[x]\,(a\,c+b\,d\,x^2)^m\,(e+f\,x)^p\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  Int[Px*(a*c+b*d*x^2)^m*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && PolyQ[Px,x] && EqQ[b*c+a*d,0] && EqQ[m,n] && (IntegerQ[m] || GtQ[a,0] && GtQ[c,0])
```

2. $\int (a+b\,x)^m\,(c+d\,x)^n\,(f\,x)^p\,\left(A+B\,x^2\right)\,dx$ when $b\,c+a\,d==0\land m==n$

1: $\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx$ when $b\,c-a\,d\neq0\land a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)==0\land m\neq-1$

Derivation: Trinomial recurrence 2b with $c=0$ and $a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)==0$

Rule 1.1.3.4.5.1: If $b\,c-a\,d\neq0\land a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)==0\land m\neq-1$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,(c+d\,x^n)\,dx\;\longrightarrow\;\frac{c\,(e\,x)^{m+1}\,(a+b\,x^n)^{p+1}}{a\,e\,(m+1)}$$

Program code:

```
Int[(e_.*x_)^m_.*(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*(e*x)^(m+1)*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(a1*a2*e*(m+1)) /;
FreeQ[{a1,b1,a2,b2,c,d,e,m,n,p},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && EqQ[a1*a2*d*(m+1)-b1*b2*c*(m+n*(p+1)+1),0] && NeQ[m
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx$ when $b\,c-a\,d\neq 0\ \land\ (n\in\mathbb{Z}\ \lor\ e>0)\ \land\ (n>0\ \land\ m<-1\ \lor\ n<0\ \land\ m+n>-1)$

Derivation: Trinomial recurrence 3b with $c = 0$

Rule 1.1.3.4.5.3: If $b\,c-a\,d\neq 0\ \land\ (n\in\mathbb{Z}\ \lor\ e>0)\ \land\ (n>0\ \land\ m<-1\ \lor\ n<0\ \land\ m+n>-1)$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx \longrightarrow \frac{c\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,e\,(m+1)}+\frac{a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)}{a\,e^n\,(m+1)}\int (e\,x)^{m+n}\left(a+b\,x^n\right)^p dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*(e*x)^(m+1)*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(a1*a2*e*(m+1)) +
  (a1*a2*d*(m+1)-b1*b2*c*(m+n*(p+1)+1))/(a1*a2*e^n*(m+1))*Int[(e*x)^(m+n)*(a1+b1*x^(n/2))^p*(a2+b2*x^(n/2))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,d,e,p},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && (IntegerQ[n] || GtQ[e,0]) &&
  (GtQ[n,0] && LtQ[m,-1] || LtQ[n,0] && GtQ[m+n,-1]) && Not[ILtQ[p,-1]]
```

3: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx$ when $b\,c-a\,d\neq 0\,\wedge\,p<-1$

Derivation: Trinomial recurrence 2b with $c = 0$

Rule 1.1.3.4.5.4.2: If $b\,c-a\,d\neq 0\,\wedge\,p<-1$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx \longrightarrow -\frac{\left(b\,c-a\,d\right)\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,b\,e\,n\,(p+1)}-\frac{a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)}{a\,b\,n\,(p+1)}\int (e\,x)^m \left(a+b\,x^n\right)^{p+1} dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  -(b1*b2*c-a1*a2*d)*(e*x)^(m+1)*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(a1*a2*b1*b2*e*n*(p+1)) -
  (a1*a2*d*(m+1)-b1*b2*c*(m+n*(p+1)+1))/(a1*a2*b1*b2*n*(p+1))*Int[(e*x)^m*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1),x] /;
FreeQ[{a1,b1,a2,b2,c,d,e,m,n},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && LtQ[p,-1] &&
  (Not[IntegerQ[p+1/2]] && NeQ[p,-5/4] || Not[RationalQ[m]] || IGtQ[n,0] && ILtQ[p+1/2,0] && LeQ[-1,m,-n*(p+1)])
```

4: $\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx$ when $b\,c-a\,d\neq 0\,\wedge\,m+n\,(p+1)+1\neq 0$

Derivation: Trinomial recurrence 2b with $c=0$ composed with binomial recurrence 1b

Rule 1.1.3.4.5.5: If $b\,c-a\,d\neq 0\,\wedge\,m+n\,(p+1)+1\neq 0$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \left(c+d\,x^n\right) dx\;\longrightarrow\;\frac{d\,(e\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{b\,e\,(m+n\,(p+1)+1)}-\frac{a\,d\,(m+1)-b\,c\,(m+n\,(p+1)+1)}{b\,(m+n\,(p+1)+1)}\int (e\,x)^m \left(a+b\,x^n\right)^p dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  d*(e*x)^(m+1)*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(b1*b2*e*(m+n*(p+1)+1)) -
  (a1*a2*d*(m+1)-b1*b2*c*(m+n*(p+1)+1))/(b1*b2*(m+n*(p+1)+1))*Int[(e*x)^m*(a1+b1*x^(n/2))^p*(a2+b2*x^(n/2))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,d,e,m,n,p},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && NeQ[m+n*(p+1)+1,0]
```

3: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $b\,c+a\,d==0\,\land\,m==n\,\land\,m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0$, then $\partial_x\frac{(a+b\,x)^m\,(c+d\,x)^m}{\left(a\,c+b\,d\,x^2\right)^m}==0$

Rule: If $b\,c+a\,d==0\,\land\,m==n\,\land\,m\notin\mathbb{Z}$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow\,\frac{(a+b\,x)^{\text{FracPart}[m]}\,(c+d\,x)^{\text{FracPart}[m]}}{\left(a\,c+b\,d\,x^2\right)^{\text{FracPart}[m]}}\int P[x]\,\left(a\,c+b\,d\,x^2\right)^m\,(e+f\,x)^p\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_.,x_Symbol] :=
  (a+b*x)^FracPart[m]*(c+d*x)^FracPart[m]/(a*c+b*d*x^2)^FracPart[m]*Int[Px*(a*c+b*d*x^2)^m*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && PolyQ[Px,x] && EqQ[b*c+a*d,0] && EqQ[m,n] && Not[IntegerQ[m]]
```

3. $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when PolynomialRemainder[P[x], a + b x, x] == 0

1: $\int \frac{P[x]\,(c+d\,x)^n\,(e+f\,x)^p}{a+b\,x}\,dx$ when PolynomialRemainder[P[x], a + b x, x] == 0

Derivation: Algebraic simplification

Basis: If PolynomialRemainder[P[x], a + b x, x] == 0, then
$\frac{P[x]}{a+b\,x}$ == PolynomialQuotient[P[x], a + b x, x]

Rule: If PolynomialRemainder[P[x], a + b x, x] == 0, then

$$\int \frac{P[x]\,(c+d\,x)^n\,(e+f\,x)^p}{a+b\,x}\,dx \longrightarrow \int \text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]\,(c+d\,x)^n\,(e+f\,x)^p\,dx$$

Program code:

```
Int[Px_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_./(a_.+b_.*x_),x_Symbol] :=
  Int[PolynomialQuotient[Px,a+b*x,x]*(c+d*x)^n*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+b*x,x],0]
```

2: $\int P[x] (a+b x)^m (c+d x)^n (e+f x)^p \, dx$ when PolynomialRemainder[P[x], a+b x, x] == 0

Derivation: Algebraic expansion

Basis: If PolynomialRemainder[P[x], a+b x, x] == 0, then
P[x] == (a+b x) PolynomialQuotient[P[x], a+b x, x]

Rule: If PolynomialRemainder[P[x], a+b x, x] == 0, then

$$\int P[x] (a+b x)^m (c+d x)^n (e+f x)^p \, dx \to \int \text{PolynomialQuotient}[P[x], a+b x, x] (a+b x)^{m+1} (c+d x)^n (e+f x)^p \, dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,a+b*x,x]*(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+b*x,x],0]
```

4: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $(m\mid n)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p,\,x\right]dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*x)^m*(c+d*x)^n*(e+f*x)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && PolyQ[Px,x] && IntegersQ[m,n]
```

5: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m<-1$

Derivation: Algebraic expansion and nondegenerate trilinear recurrence 3

Basis: Let $Q[x]\to\text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]$ and $R\to\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x]$, then $P[x]==Q[x]\,(a+b\,x)+R$

Note: If the integrand has a negative integer exponent, incrementing it, rather than another negative fractional exponent, produces simpler antiderivatives.

Rule: If $m<-1$, let $Q[x]\to\text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]$ and $R\to\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x]$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow$$

$$\int Q[x]\,(a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\,dx+R\int(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\;\longrightarrow$$

$$\frac{b\,R\,(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}+\frac{1}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}\int(a+b\,x)^{m+1}\,(c+d\,x)^{n}\,(e+f\,x)^{p}\cdot\left((m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)\,Q[x]+a\,d\,f\,R\,(m+1)-b\,R\,(d\,e\,(m+n+2)+c\,f\,(m+p+2))-b\,d\,f\,R\,(m+n+p+3)\,x\right)dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[Px,a+b*x,x], R=PolynomialRemainder[Px,a+b*x,x]},
  b*R*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    ExpandToSum[(m+1)*(b*c-a*d)*(b*e-a*f)*Qx+a*d*f*R*(m+1)-b*R*(d*e*(m+n+2)+c*f*(m+p+2))-b*d*f*R*(m+n+p+3)*x,x],x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && PolyQ[Px,x] && ILtQ[m,-1] && IntegersQ[2*m,2*n,2*p]
```

```
Int[Px_*(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[Px,a+b*x,x], R=PolynomialRemainder[Px,a+b*x,x]},
  b*R*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    ExpandToSum[(m+1)*(b*c-a*d)*(b*e-a*f)*Qx+a*d*f*R*(m+1)-b*R*(d*e*(m+n+2)+c*f*(m+p+2))-b*d*f*R*(m+n+p+3)*x,x],x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && PolyQ[Px,x] && LtQ[m,-1] && IntegersQ[2*m,2*n,2*p]
```

6: $\int P_q[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$ when $m+q+n+p+1\neq 0$

Derivation: Algebraic expansion and nondegenerate trilinear recurrence 2

Rule: If $m+q+n+p+1\neq 0$, then

$$\int P_q[x]\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow$$

$$\int\left(P_q[x]-\frac{P_q[x,\,q]}{b^q}\,(a+b\,x)^q\right)(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx+\frac{P_q[x,\,q]}{b^q}\int(a+b\,x)^{m+q}\,(c+d\,x)^n\,(e+f\,x)^p\,dx\,\longrightarrow$$

$$\frac{P_q[x,q]\,(a+b\,x)^{m+q-1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{d\,f\,b^{q-1}\,(m+q+n+p+1)}+\frac{1}{d\,f\,b^{q}\,(m+q+n+p+1)}\int(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\cdot\\
\Big(d\,f\,b^q\,(m+q+n+p+1)\,P_q[x]-d\,f\,P_q[x,q]\,(m+q+n+p+1)\,(a+b\,x)^q+\\
P_q[x,q]\,(a+b\,x)^{q-2}\,\big(a^2\,d\,f\,(m+q+n+p+1)-b\,(b\,c\,e\,(m+q-1)+a\,(d\,e\,(n+1)+c\,f\,(p+1)))+\\
b\,(a\,d\,f\,(2\,(m+q)+n+p)-b\,(d\,e\,(m+q+n)+c\,f\,(m+q+p)))\,x\big)\Big)\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.,x_Symbol] :=
  With[{q=Expon[Px,x],k=Coeff[Px,x,Expon[Px,x]]},
  k*(a+b*x)^(m+q-1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*b^(q-1)*(m+q+n+p+1)) +
  1/(d*f*b^q*(m+q+n+p+1))*Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*
    ExpandToSum[d*f*b^q*(m+q+n+p+1)*Px-d*f*k*(m+q+n+p+1)*(a+b*x)^q +
      k*(a+b*x)^(q-2)*(a^2*d*f*(m+q+n+p+1)-b*(b*c*e*(m+q-1)+a*(d*e*(n+1)+c*f*(p+1)))+
        b*(a*d*f*(2*(m+q)+n+p)-b*(d*e*(m+q+n)+c*f*(m+q+p)))*x),x],x] /;
  NeQ[m+q+n+p+1,0]] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && PolyQ[Px,x] && IntegersQ[2*m,2*n,2*p]
```

Rules for integrands of the form $P[x]\ (a+b\,x)^m\ (c+d\,x)^n$

1. $\int P[x]\ (a+b\,x)^m\ (c+d\,x)^n\,dx$ when $b\,c+a\,d==0\ \land\ m==n$

   1: $\int P[x]\ (a+b\,x)^m\ (c+d\,x)^n\,dx$ when $b\,c+a\,d==0\ \land\ m==n\ \land\ (m\in\mathbb{Z}\ \lor\ a>0\ \land\ c>0)$

Derivation: Algebraic simplification

Basis: If $b\,c+a\,d==0\ \land\ (m\in\mathbb{Z}\ \lor\ a>0\ \land\ c>0)$, then $(a+b\,x)^m\ (c+d\,x)^m==\left(a\,c+b\,d\,x^2\right)^m$

Rule: If $b\,c+a\,d==0\ \land\ m==n\ \land\ (m\in\mathbb{Z}\ \lor\ a>0\ \land\ c>0)$, then

$$\int P[x]\ (a+b\,x)^m\ (c+d\,x)^n\,dx\ \longrightarrow\ \int P[x]\ \left(a\,c+b\,d\,x^2\right)^m\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.,x_Symbol] :=
  Int[Px*(a*c+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d,m,n},x] && PolyQ[Px,x] && EqQ[b*c+a*d,0] && EqQ[m,n] && (IntegerQ[m] || GtQ[a,0] && GtQ[c,0])
```

2. $\int (a+b\,x)^m\ (c+d\,x)^n\ \left(A+B\,x^2\right)\,dx$ when $b\,c+a\,d==0\ \land\ m==n$

   1: $\int \left(a+b\,x^n\right)^p\ \left(c+d\,x^n\right)\,dx$ when $b\,c-a\,d\neq 0\ \land\ a\,d-b\,c\ (n\ (p+1)+1)==0$

Derivation: Trinomial recurrence 2b with $c=0$, $p=0$ and $a\,d-b\,c\ (n\ (p+1)+1)==0$

Rule 1.1.3.3.7.1: If $b\,c-a\,d\neq 0\ \land\ a\,d-b\,c\ (n\ (p+1)+1)==0$, then

$$\int \left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)\,dx \;\longrightarrow\; \frac{c\,x\,\left(a+b\,x^n\right)^{p+1}}{a}$$

Program code:

```
Int[(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*x*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(a1*a2) /;
FreeQ[{a1,b1,a2,b2,c,d,n,p},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && EqQ[a1*a2*d-b1*b2*c*(n*(p+1)+1),0]
```

2: $\int \left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,p<-1$

Derivation: Trinomial recurrence 2b with $c=0$ and $p=0$

Rule 1.1.3.3.7.2: If $b\,c-a\,d\neq 0\,\land\,p<-1$, then

$$\int \left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)\,dx \;\longrightarrow\; -\frac{\left(b\,c-a\,d\right)\,x\,\left(a+b\,x^n\right)^{p+1}}{a\,b\,n\,(p+1)}-\frac{a\,d-b\,c\,(n\,(p+1)+1)}{a\,b\,n\,(p+1)}\int \left(a+b\,x^n\right)^{p+1}\,dx$$

Program code:

```
Int[(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  -(b1*b2*c-a1*a2*d)*x*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(a1*a2*b1*b2*n*(p+1)) -
  (a1*a2*d-b1*b2*c*(n*(p+1)+1))/(a1*a2*b1*b2*n*(p+1))*Int[(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1),x] /;
FreeQ[{a1,b1,a2,b2,c,d,n},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && (LtQ[p,-1] || ILtQ[1/n+p,0])
```

3: $\int \left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\,(p+1)+1\neq 0$

Derivation: Trinomial recurrence 2b with $c=0$ and $p=0$ composed with binomial recurrence 1b with $p=0$

Rule 1.1.3.3.7.4: If $b\,c-a\,d\neq 0\,\land\,n\,(p+1)+1\neq 0$, then

$$\int (a+b x^n)^p (c+d x^n)\, dx \longrightarrow \frac{d x (a+b x^n)^{p+1}}{b (n (p+1)+1)} - \frac{a d - b c (n (p+1)+1)}{b (n (p+1)+1)} \int (a+b x^n)^p\, dx$$

Program code:

```
Int[(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  d*x*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(b1*b2*(n*(p+1)+1)) -
  (a1*a2*d-b1*b2*c*(n*(p+1)+1))/(b1*b2*(n*(p+1)+1))*Int[(a1+b1*x^(n/2))^p*(a2+b2*x^(n/2))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,d,n,p},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && NeQ[n*(p+1)+1,0]
```

3: $\int P[x] (a+b x)^m (c+d x)^n\, dx$ when $b c + a d == 0 \land m == n \land m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b c + a d == 0$, then $\partial_x \frac{(a+b x)^m (c+d x)^m}{(a c+b d x^2)^m} == 0$

Rule: If $b c + a d == 0 \land m == n \land m \notin \mathbb{Z}$, then

$$\int P[x] (a+b x)^m (c+d x)^n\, dx \longrightarrow \frac{(a+b x)^{\text{FracPart}[m]} (c+d x)^{\text{FracPart}[m]}}{(a c+b d x^2)^{\text{FracPart}[m]}} \int P[x] (a c+b d x^2)^m\, dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^FracPart[m]*(c+d*x)^FracPart[m]/(a*c+b*d*x^2)^FracPart[m]*Int[Px*(a*c+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d,m,n},x] && PolyQ[Px,x] && EqQ[b*c+a*d,0] && EqQ[m,n] && Not[IntegerQ[m]]
```

2. $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx$ when $\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x]==0$

1: $\int \frac{P[x]\,(c+d\,x)^n}{a+b\,x}\,dx$ when $\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x]==0$

Derivation: Algebraic simplification

Basis: If $\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x]==0$, then
$\frac{P[x]}{a+b\,x}==\text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]$

Rule: If $\text{PolynomialRemainder}[P[x],\,a+b\,x,\,x]==0$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx\;\longrightarrow\;\int \text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]\,(c+d\,x)^n\,dx$$

Program code:

```
Int[Px_*(c_.+d_.*x_)^n_./(a_.+b_.*x_),x_Symbol] :=
  Int[PolynomialQuotient[Px,a+b*x,x]*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,n},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+b*x,x],0]
```

2: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx$ when PolynomialRemainder[P[x], a + b x, x] == 0

Derivation: Algebraic expansion

Basis: If PolynomialRemainder[P[x], a + b x, x] == 0, then
P[x] == (a + b x) PolynomialQuotient[P[x], a + b x, x]

Rule: If PolynomialRemainder[P[x], a + b x, x] == 0, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow \int \text{PolynomialQuotient}[P[x],\,a+b\,x,\,x]\,(a+b\,x)^{m+1}\,(c+d\,x)^n\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.,x_Symbol] :=
  Int[PolynomialQuotient[Px,a+b*x,x]*(a+b*x)^(m+1)*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,m,n},x] && PolyQ[Px,x] && EqQ[PolynomialRemainder[Px,a+b*x,x],0]
```

3: $\int \frac{P[x]\,(c+d\,x)^n}{a+b\,x}\,dx$ when $n+\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Rule: If $n+\frac{1}{2}\in\mathbb{Z}^-$, then

$$\int \frac{P[x]\,(c+d\,x)^n}{a+b\,x}\,dx \,\longrightarrow\, \int \frac{1}{\sqrt{c+d\,x}}\,\text{ExpandIntegrand}\left[\frac{P[x]\,(c+d\,x)^{n+\frac{1}{2}}}{a+b\,x},\,x\right]dx$$

Program code:

```
Int[Px_*(c_.+d_.*x_)^n_./(a_.+b_.*x_),x_Symbol] :=
  Int[ExpandIntegrand[1/Sqrt[c+d*x],Px*(c+d*x)^(n+1/2)/(a+b*x),x],x] /;
FreeQ[{a,b,c,d,n},x] && PolyQ[Px,x] && ILtQ[n+1/2,0] && GtQ[Expon[Px,x],2]
```

4: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx$ when $(m\mid n)\in\mathbb{Z}\,\lor\,m+2\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}\,\lor\,m+2\in\mathbb{Z}^+$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx \,\longrightarrow\, \int \text{ExpandIntegrand}\left[P[x]\,(a+b\,x)^m\,(c+d\,x)^n,\,x\right]dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*x)^m*(c+d*x)^n,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && PolyQ[Px,x] && (IntegersQ[m,n] || IGtQ[m,-2]) && GtQ[Expon[Px,x],2]
```

5: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx$ when $m<-1$

Derivation: Algebraic expansion and linear recurrence 3

Basis: Let $Q[x] \to \text{PolynomialQuotient}[P[x],\ a+b\,x,\ x]$ and $R \to \text{PolynomialRemainder}[P[x],\ a+b\,x,\ x]$, then $P[x] == Q[x]\,(a+b\,x)+R$

Note: If the integrand has a negative integer exponent, incrementing it, rather than another negative fractional exponent, produces simpler antiderivatives.

Rule: If $m<-1$, let $Q[x] \to \text{PolynomialQuotient}[P[x],\ a+b\,x,\ x]$ and $R \to \text{PolynomialRemainder}[P[x],\ a+b\,x,\ x]$, then

$$\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow$$

$$\int Q[x]\,(a+b\,x)^{m+1}\,(c+d\,x)^n\,dx+R\int (a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow$$

$$\frac{R\,(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(m+1)\,(b\,c-a\,d)}+\frac{1}{(m+1)\,(b\,c-a\,d)}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,\left((m+1)\,(b\,c-a\,d)\,Q[x]-d\,R\,(m+n+2)\right)\,dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[Px,a+b*x,x], R=PolynomialRemainder[Px,a+b*x,x]},
  R*(a+b*x)^(m+1)*(c+d*x)^(n+1)/((m+1)*(b*c-a*d)) +
  1/((m+1)*(b*c-a*d))*Int[(a+b*x)^(m+1)*(c+d*x)^n*ExpandToSum[(m+1)*(b*c-a*d)*Qx-d*R*(m+n+2),x],x]] /;
FreeQ[{a,b,c,d,n},x] && PolyQ[Px,x] && ILtQ[m,-1] && GtQ[Expon[Px,x],2]
```

```
Int[Px_*(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[Px,a+b*x,x], R=PolynomialRemainder[Px,a+b*x,x]},
  R*(a+b*x)^(m+1)*(c+d*x)^(n+1)/((m+1)*(b*c-a*d)) +
  1/((m+1)*(b*c-a*d))*Int[(a+b*x)^(m+1)*(c+d*x)^n*ExpandToSum[(m+1)*(b*c-a*d)*Qx-d*R*(m+n+2),x],x]] /;
FreeQ[{a,b,c,d,n},x] && PolyQ[Px,x] && LtQ[m,-1] && GtQ[Expon[Px,x],2]
```

6: $\int P[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx$ when $m+q+n+1\neq 0$

Derivation: Algebraic expansion and linear recurrence 2

Rule: If $m+q+n+1\neq 0$, then

$$\int P_q[x]\,(a+b\,x)^m\,(c+d\,x)^n\,dx\ \longrightarrow$$

$$\int\left(P_q[x]-\frac{P_q[x,\,q]}{b^q}\,(a+b\,x)^q\right)(a+b\,x)^m\,(c+d\,x)^n\,dx+\frac{P_q[x,\,q]}{b^q}\int(a+b\,x)^{m+q}\,(c+d\,x)^n\,dx\ \longrightarrow$$

$$\frac{P_q[x,\,q]\,(a+b\,x)^{m+q}\,(c+d\,x)^{n+1}}{d\,b^q\,(m+q+n+1)}+\frac{1}{d\,b^q\,(m+q+n+1)}\int(a+b\,x)^m\,(c+d\,x)^n\cdot$$
$$\left(d\,b^q\,(m+q+n+1)\,P_q[x]-d\,P_q[x,\,q]\,(m+q+n+1)\,(a+b\,x)^q-P_q[x,\,q]\,(b\,c-a\,d)\,(m+q)\,(a+b\,x)^{q-1}\right)dx$$

Program code:

```
Int[Px_*(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.,x_Symbol] :=
  With[{q=Expon[Px,x],k=Coeff[Px,x,Expon[Px,x]]},
  k*(a+b*x)^(m+q)*(c+d*x)^(n+1)/(d*b^q*(m+q+n+1)) +
  1/(d*b^q*(m+q+n+1))*Int[(a+b*x)^m*(c+d*x)^n*
    ExpandToSum[d*b^q*(m+q+n+1)*Px-d*k*(m+q+n+1)*(a+b*x)^q-k*(b*c-a*d)*(m+q)*(a+b*x)^(q-1),x],x] /;
  NeQ[m+q+n+1,0]] /;
FreeQ[{a,b,c,d,m,n},x] && PolyQ[Px,x] && GtQ[Expon[Px,x],2]
```