

Rules for integrands of the form $P_q[x] (a + b x^2 + c x^4)^p$

1: $\int P_q[x] (a + b x^2 + c x^4)^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.2.5.1: If $p \in \mathbb{Z}^+$, then

$$\int P_q[x] (a + b x^2 + c x^4)^p dx \rightarrow \int \text{ExpandIntegrand}[P_q[x] (a + b x^2 + c x^4)^p, x] dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && IGtQ[p,0]
```

2: $\int P_q[x] (a + b x^2 + c x^4)^p dx$ when $P_q[x, 0] == 0$

Derivation: Algebraic simplification

Rule 1.2.2.5.2: If $P_q[x, 0] == 0$, then

$$\int P_q[x] (a + b x^2 + c x^4)^p dx \rightarrow \int x \text{PolynomialQuotient}[P_q[x], x, x] (a + b x^2 + c x^4)^p dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pq,x,x]*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && EqQ[Coeff[Pq,x,0],0] && Not[MatchQ[Pq,x^m_.*u_./; IntegerQ[m]]]
```

3: $\int P_q[x] (a + bx^2 + cx^4)^p dx$ when $\neg P_q[x^2]$

Derivation: Algebraic expansion

Basis: $P_q[x] = \sum_{k=0}^{q/2} P_q[x, 2k] x^{2k} + x \sum_{k=0}^{(q-1)/2} P_q[x, 2k+1] x^{2k}$

Note: This rule transforms $P_q[x]$ into a sum of the form $Q_r[x^2] + x R_s[x^2]$.

Rule 1.2.2.5.3: If $\neg P_q[x^2]$, then

$$\int P_q[x] (a + bx^2 + cx^4)^p dx \rightarrow \int \left(\sum_{k=0}^{q/2} P_q[x, 2k] x^{2k} \right) (a + bx^2 + cx^4)^p dx + \int x \left(\sum_{k=0}^{(q-1)/2} P_q[x, 2k+1] x^{2k} \right) (a + bx^2 + cx^4)^p dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
Module[{q=Expon[Pq,x],k},
Int[Sum[Coeff[Pq,x,2*k]*x^(2*k),{k,0,q/2}]*(a+b*x^2+c*x^4)^p,x] +
Int[x*Sum[Coeff[Pq,x,2*k+1]*x^(2*k),{k,0,(q-1)/2}]*(a+b*x^2+c*x^4)^p,x]] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && Not[PolyQ[Pq,x^2]]
```

4: $\int (d + ex^2 + fx^4) (a + bx^2 + cx^4)^p dx$ when $a e - b d (2p + 3) == 0 \wedge a f - c d (4p + 5) == 0$

Rule 1.2.2.5.4: If $a e - b d (2p + 3) == 0 \wedge a f - c d (4p + 5) == 0$, then

$$\int (d + ex^2 + fx^4) (a + bx^2 + cx^4)^p dx \rightarrow \frac{dx (a + bx^2 + cx^4)^{p+1}}{a}$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{d=Coeff[Pq,x,0],e=Coeff[Pq,x,2],f=Coeff[Pq,x,4]},
    d*x*(a+b*x^2+c*x^4)^(p+1)/a /;
    EqQ[a*e-b*d*(2*p+3),0] && EqQ[a*f-c*d*(4*p+5),0] /;
    FreeQ[{a,b,c,p},x] && PolyQ[Pq,x^2] && EqQ[Expon[Pq,x],4]
```

5: $\int (d + ex^2 + fx^4 + gx^6) (a + bx^2 + cx^4)^p dx$ when $3a^2g - c(4p+7)(ae - bd(2p+3)) = 0 \wedge 3a^2f - 3acd(4p+5) - b(2p+5)(ae - bd(2p+3)) = 0$

Rule 1.2.2.5.5: If $3a^2g - c(4p+7)(ae - bd(2p+3)) = 0 \wedge$, then
 $3a^2f - 3acd(4p+5) - b(2p+5)(ae - bd(2p+3)) = 0$

$$\int (d + ex^2 + fx^4 + gx^6) (a + bx^2 + cx^4)^p dx \rightarrow \frac{x(3ad + (ae - bd(2p+3))x^2)(a + bx^2 + cx^4)^{p+1}}{3a^2}$$

Program code:

```
Int[Pq*(a+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  With[{d=Coeff[Pq,x,0],e=Coeff[Pq,x,2],f=Coeff[Pq,x,4],g=Coeff[Pq,x,6]},
    x*(3*a*d+(a*e-b*d*(2*p+3))*x^2)*(a+b*x^2+c*x^4)^(p+1)/(3*a^2) /;
    EqQ[3*a^2*g-c*(4*p+7)*(a*e-b*d*(2*p+3)),0] && EqQ[3*a^2*f-3*a*c*d*(4*p+5)-b*(2*p+5)*(a*e-b*d*(2*p+3)),0] /;
    FreeQ[{a,b,c,p},x] && PolyQ[Pq,x^2] && EqQ[Expon[Pq,x],6]
```

6: $\int \frac{P_q[x^2]}{a + b x^2 + c x^4} dx$ when $q > 1$

Derivation: Algebraic expansion

Rule 1.2.2.5.6: If $q > 1$, then

$$\int \frac{P_q[x^2]}{a + b x^2 + c x^4} dx \rightarrow \int \text{ExpandIntegrand}\left[\frac{P_q[x^2]}{a + b x^2 + c x^4}, x\right] dx$$

Program code:

```
Int[Pq_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[Pq/(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x^2] && Expon[Pq,x^2]>1
```

7: $\int P_q[x^2] (a+bx^2+cx^4)^p dx$ when $q > 1 \wedge b^2 - 4ac = 0$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4ac = 0$, then $\partial_x \frac{(a+bx^2+cx^4)^p}{(b+2cx^2)^{2p}} = 0$

Rule 1.2.2.5.7: If $q > 1 \wedge b^2 - 4ac = 0$, then

$$\int P_q[x^2] (a+bx^2+cx^4)^p dx \rightarrow \frac{(a+bx^2+cx^4)^{\text{FracPart}[p]}}{(4c)^{\text{IntPart}[p]} (b+2cx^2)^{2\text{FracPart}[p]}} \int P_q[x^2] (b+2cx^2)^{2p} dx$$

Program code:

```
Int[Pq*(a+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x^2)^(2*FracPart[p]))*Int[Pq*(b+2*c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x^2] && Expon[Pq,x^2]>1 && EqQ[b^2-4*a*c,0]
```

8. $\int P_q[x^2] (a+bx^2+cx^4)^p dx$ when $q > 1 \wedge b^2 - 4ac \neq 0$

1: $\int P_q[x^2] (a+bx^2+cx^4)^p dx$ when $q > 1 \wedge b^2 - 4ac \neq 0 \wedge p < -1$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.2.5.8.1: If $q > 1 \wedge b^2 - 4ac \neq 0 \wedge p < -1$, let

$Q_{q-2}[x^2] \rightarrow \text{PolynomialQuotient}[P_q[x^2], a+bx^2+cx^4, x]$ and

$d+ex^2 \rightarrow \text{PolynomialRemainder}[P_q[x^2], a+bx^2+cx^4, x]$, then

$$\int P_q[x^2] (a+bx^2+cx^4)^p dx \rightarrow$$

$$\int (d+ex^2) (a+bx^2+cx^4)^p dx + \int Q_{q-2}[x^2] (a+bx^2+cx^4)^{p+1} dx \rightarrow$$

$$\frac{x (a+bx^2+cx^4)^{p+1} (abe - d(b^2 - 2ac) - c(bd - 2ae)x^2)}{2a(p+1)(b^2 - 4ac)} +$$

$$\frac{1}{2a(p+1)(b^2 - 4ac)} \int (a+bx^2+cx^4)^{p+1} (2a(p+1)(b^2 - 4ac) Q_{q-2}[x^2] + b^2d(2p+3) - 2acd(4p+5) - abe + c(4p+7)(bd - 2ae)x^2) dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{d=Coeff[PolynomialRemainder[Pq,a+b*x^2+c*x^4,x],x,0],
    e=Coeff[PolynomialRemainder[Pq,a+b*x^2+c*x^4,x],x,2]},
    x*(a+b*x^2+c*x^4)^(p+1)*(a*b*e-d*(b^2-2*a*c)-c*(b*d-2*a*e)*x^2)/(2*a*(p+1)*(b^2-4*a*c)) +
    1/(2*a*(p+1)*(b^2-4*a*c))*Int[(a+b*x^2+c*x^4)^(p+1)*
      ExpandToSum[2*a*(p+1)*(b^2-4*a*c)*PolynomialQuotient[Pq,a+b*x^2+c*x^4,x]+
        b^2*d*(2*p+3)-2*a*c*d*(4*p+5)-a*b*e+c*(4*p+7)*(b*d-2*a*e)*x^2,x],x] /;
    FreeQ[{a,b,c},x] && PolyQ[Pq,x^2] && Expon[Pq,x^2]>1 && NeQ[b^2-4*a*c,0] && LtQ[p,-1]
```

2: $\int P_q[x^2] (a+bx^2+cx^4)^p dx$ when $q > 1 \wedge b^2 - 4ac \neq 0 \wedge p \neq -1$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A = 0, B = 1$ and $m = m - n$

Reference: G&R 2.104

Note: If $q \geq 2 \wedge p \neq -1$, then $2q + 4p + 1 \neq 0$.

Rule 1.2.2.5.8.2: If $q > 1 \wedge b^2 - 4ac \neq 0 \wedge p \neq -1$, let $e \rightarrow P_q[x^2, q]$, then

$$\int P_q[x^2] (a+bx^2+cx^4)^p dx \rightarrow$$

$$\int (P_q[x^2] - ex^{2q}) (a+bx^2+cx^4)^p dx + e \int x^{2q} (a+bx^2+cx^4)^p dx \rightarrow$$

$$\frac{e x^{2q-3} (a + b x^2 + c x^4)^{p+1}}{c (2q + 4p + 1)} + \frac{1}{c (2q + 4p + 1)} \int (a + b x^2 + c x^4)^p .$$

$$(c (2q + 4p + 1) P_q[x^2] - a e (2q - 3) x^{2q-4} - b e (2q + 2p - 1) x^{2q-2} - c e (2q + 4p + 1) x^{2q}) dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x^2],e=Coeff[Pq,x^2,Expon[Pq,x^2]]},
    e*x^(2*q-3)*(a+b*x^2+c*x^4)^(p+1)/(c*(2*q+4*p+1)) +
    1/(c*(2*q+4*p+1))*Int[(a+b*x^2+c*x^4)^p*
      ExpandToSum[c*(2*q+4*p+1)*Pq-a*e*(2*q-3)*x^(2*q-4)-b*e*(2*q+2*p-1)*x^(2*q-2)-c*e*(2*q+4*p+1)*x^(2*q),x],x] /;
    FreeQ[{a,b,c,p},x] && PolyQ[Pq,x^2] && Expon[Pq,x^2]>1 && NeQ[b^2-4*a*c,0] && Not[LtQ[p,-1]]
```


S: $\int P_q[x] (a + bx + cx^2 + dx^3 + ex^4)^p dx$ when $d^3 - 4cde + 8be^2 = 0 \wedge p \notin \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $d^3 - 4cde + 8be^2 = 0$, then

$$(a + bx + cx^2 + dx^3 + ex^4)^p =$$

$$\text{Subst} \left[\left(a + \frac{d^4}{256e^3} - \frac{bd}{8e} + \left(c - \frac{3d^2}{8e} \right) x^2 + ex^4 \right)^p, x, \frac{d}{4e} + x \right] \partial_x \left(\frac{d}{4e} + x \right)$$

Rule: If $d^3 - 4cde + 8be^2 = 0 \wedge p \notin \mathbb{Z}^+$, then

$$\int P_q[x] (a + bx + cx^2 + dx^3 + ex^4)^p dx \rightarrow \text{Subst} \left[\int P_q \left[x - \frac{d}{4e} \right] \left(a + \frac{d^4}{256e^3} - \frac{bd}{8e} + \left(c - \frac{3d^2}{8e} \right) x^2 + ex^4 \right)^p dx, x, \frac{d}{4e} + x \right]$$

Program code:

```
Int[Pq_*Q4_^p_,x_Symbol] :=
  With[{a=Coeff[Q4,x,0],b=Coeff[Q4,x,1],c=Coeff[Q4,x,2],d=Coeff[Q4,x,3],e=Coeff[Q4,x,4]},
    Subst[Int[SimplifyIntegrand[ReplaceAll[Pq,x->-d/(4*e)+x]*(a+d^4/(256*e^3)-b*d/(8*e)+(c-3*d^2/(8*e))*x^2+e*x^4)^p,x],x],x,d/(4*e)
    EqQ[d^3-4*c*d*e+8*b*e^2,0] && NeQ[d,0]] /;
  FreeQ[p,x] && PolyQ[Pq,x] && PolyQ[Q4,x,4] && Not[IGtQ[p,0]]
```