Rules for integrands of the form $u\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p$

1: $\int (a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx$

Derivation: Integration by substitution

Rule:

$$\int (a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx \longrightarrow \frac{1}{e}\,\mathrm{Subst}\Big[\int (a+b\,\mathrm{Log}[c\,x^n])^p\,dx,\ x,\ d+e\,x\Big]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  1/e*Subst[Int[(a+b*Log[c*x^n])^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,n,p},x]
```

2. $\int (f+g\,x^r)^q\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx$

   1. $\int (f+g\,x)^q\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx$

      1: $\int (f+g\,x)^q\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g=0$

Derivation: Integration by substitution

Basis: If $e\,f-d\,g=0$, then $(f+g\,x)^q\,F[d+e\,x]=\frac{1}{e}\,\mathrm{Subst}\left[\left(\frac{f\,x}{d}\right)^q F[x],\,x,\,d+e\,x\right]\partial_x\,(d+e\,x)$

Rule: If $e\,f-d\,g=0$, then

$$\int (f+g\,x)^q\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx\;\to\;\frac{1}{e}\,\mathrm{Subst}\left[\int\left(\frac{f\,x}{d}\right)^q(a+b\,\mathrm{Log}[c\,x^n])^p\,dx,\,x,\,d+e\,x\right]$$

Program code:

```
Int[(f_+g_.x_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  1/e*Subst[Int[(f*x/d)^q*(a+b*Log[c*x^n])^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q},x] && EqQ[e*f-d*g,0]
```

      2. $\int (f+g\,x)^q\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g\neq 0$

         1. $\int (f+g\,x)^q\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,p>0$

            1. $\int (f+g\,x)^q\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])\,dx$ when $e\,f-d\,g\neq 0$

               1. $\int \frac{(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])}{f+g\,x}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,p\in\mathbb{Z}^+$

                  1. $\int \frac{a+b\,\mathrm{Log}[c\,(d+e\,x)]}{x}\,dx$ when $c\,d>0$

1: $\int \frac{\text{Log}[c\,(d+e\,x^n)]}{x}\,dx$ when $c\,d == 1$

Rule: If $c\,d == 1$, then

$$\int \frac{\text{Log}[c\,(d+e\,x^n)]}{x}\,dx \longrightarrow -\frac{\text{PolyLog}[2,\,-c\,e\,x^n]}{n}$$

Program code:

```
Int[Log[c_.*(d_+e_.*x_^n_.)]/x_,x_Symbol] :=
  -PolyLog[2,-c*e*x^n]/n /;
FreeQ[{c,d,e,n},x] && EqQ[c*d,1]
```

2: $\int \frac{a+b\,\text{Log}[c\,(d+e\,x)]}{x}\,dx$ when $c\,d > 0$

Derivation: Algebraic expansion

Basis: If $c\,d > 0$, then $\text{Log}[c\,(d+e\,x)] == \text{Log}[c\,d] + \text{Log}\left[1+\frac{e\,x}{d}\right]$

Rule: If $c\,d > 0$, then

$$\int \frac{a+b\,\text{Log}[c\,(d+e\,x)]}{x}\,dx \longrightarrow (a+b\,\text{Log}[c\,d])\,\text{Log}[x] + b\int \frac{\text{Log}\left[1+\frac{e\,x}{d}\right]}{x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)])/x_,x_Symbol] :=
  (a+b*Log[c*d])*Log[x] + b*Int[Log[1+e*x/d]/x,x] /;
FreeQ[{a,b,c,d,e},x] && GtQ[c*d,0]
```

2: $\int \frac{a+b\,\text{Log}[c\,(d+e\,x)]}{f+g\,x}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,g+c\,(e\,f-d\,g)==0$

Derivation: Integration by substitution

Basis: If $g+c\,(e\,f-d\,g)==0$, then $F[c\,(d+e\,x)]==\frac{1}{g}\,\text{Subst}\left[F\left[1+\frac{c\,e\,x}{g}\right],\,x,\,f+g\,x\right]\,\partial_x\,(f+g\,x)$

Rule: If $e\,f-d\,g\neq 0\,\wedge\,g+c\,(e\,f-d\,g)==0$, then

$$\int \frac{a+b\,\text{Log}[c\,(d+e\,x)]}{f+g\,x}\,dx\;\rightarrow\;\frac{1}{g}\,\text{Subst}\left[\int \frac{a+b\,\text{Log}\left[1+\frac{c\,e\,x}{g}\right]}{x}\,dx,\,x,\,f+g\,x\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)])/(f_.+g_.x_),x_Symbol] :=
  1/g*Subst[Int[(a+b*Log[1+c*e*x/g])/x,x],x,f+g*x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[e*f-d*g,0] && EqQ[g+c*(e*f-d*g),0]
```

3: $\int \frac{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]}{f+g\,x}\,dx$ when $e\,f-d\,g\neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{f+g\,x} == \frac{1}{g}\,\partial_x\,\text{Log}\left[\frac{e\,(f+g\,x)}{e\,f-d\,g}\right]$

Rule: If $e\,f-d\,g\neq 0$, then

$$\int \frac{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]}{f+g\,x}\,dx \longrightarrow \frac{\text{Log}\left[\frac{e\,(f+g\,x)}{e\,f-d\,g}\right]\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)}{g}-\frac{b\,e\,n}{g}\int \frac{\text{Log}\left[\frac{e\,(f+g\,x)}{e\,f-d\,g}\right]}{d+e\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/(f_.+g_.x_),x_Symbol] :=
  Log[e*(f+g*x)/(e*f-d*g)]*(a+b*Log[c*(d+e*x)^n])/g - b*e*n/g*Int[Log[(e*(f+g*x))/(e*f-d*g)]/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0]
```

2: $\int (f+g\,x)^q\,\left(a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]\right)\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,q\neq -1$

Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'

Derivation: Integration by parts

Basis: $(f+g\,x)^q = \partial_x\,\frac{(f+g\,x)^{q+1}}{g\,(q+1)}$

Rule: If $e\,f-d\,g\neq 0\,\wedge\,q\neq -1$, then

$$\int (f+g\,x)^q\,\left(a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]\right)\,dx \,\longrightarrow\, \frac{(f+g\,x)^{q+1}\,\left(a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]\right)}{g\,(q+1)}-\frac{b\,e\,n}{g\,(q+1)}\int \frac{(f+g\,x)^{q+1}}{d+e\,x}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
  (f+g*x)^(q+1)*(a+b*Log[c*(d+e*x)^n])/(g*(q+1)) -
  b*e*n/(g*(q+1))*Int[(f+g*x)^(q+1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n,q},x] && NeQ[e*f-d*g,0] && NeQ[q,-1]
```

2: $\int \frac{\left(a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]\right)^p}{f+g\,x}\,dx$ when $e\,f-d\,g\neq 0\,\wedge\,p-1\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\frac{1}{f+g\,x} = \frac{1}{g}\,\partial_x\,\mathrm{Log}\left[\frac{e\,(f+g\,x)}{e\,f-d\,g}\right]$

Basis: $\partial_x\,\left(a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]\right)^p = \frac{b\,e\,n\,p\,\left(a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]\right)^{p-1}}{d+e\,x}$

Rule: If $e\,f-d\,g\neq 0\,\wedge\,p-1\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p}{f+g\,x}\,dx \to \frac{\mathrm{Log}\left[\frac{e\,(f+g\,x)}{e\,f-d\,g}\right]\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p}{g}-\frac{b\,e\,n\,p}{g}\int \frac{\mathrm{Log}\left[\frac{e\,(f+g\,x)}{e\,f-d\,g}\right]\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^{p-1}}{d+e\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_/(f_.+g_.x_),x_Symbol] :=
  Log[e*(f+g*x)/(e*f-d*g)]*(a+b*Log[c*(d+e*x)^n])^p/g -
  b*e*n*p/g*Int[Log[(e*(f+g*x))/(e*f-d*g)]*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && IGtQ[p,1]
```

3: $\int \frac{(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p}{(f+g\,x)^2}\,dx$ when $e\,f-d\,g\neq 0 \land p>0$

Derivation: Integration by parts

Basis: $\frac{1}{(f+g\,x)^2} = \partial_x \frac{d+e\,x}{(e\,f-d\,g)\,(f+g\,x)}$

Rule: If $e\,f-d\,g\neq 0 \land p>0$, then

$$\int \frac{(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p}{(f+g\,x)^2}\,dx \to \frac{(d+e\,x)\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p}{(e\,f-d\,g)\,(f+g\,x)}-\frac{b\,e\,n\,p}{e\,f-d\,g}\int \frac{(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^{p-1}}{f+g\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_/(f_.+g_.*x_)^2,x_Symbol] :=
  (d+e*x)*(a+b*Log[c*(d+e*x)^n])^p/((e*f-d*g)*(f+g*x)) -
  b*e*n*p/(e*f-d*g)*Int[(a+b*Log[c*(d+e*x)^n])^(p-1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && GtQ[p,0]
```

4: $\int (f+g\,x)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,p>0\,\land\,q\neq -1$

Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'

Derivation: Integration by parts

Basis: $(f+g\,x)^q = \partial_x \frac{(f+g\,x)^{q+1}}{g\,(q+1)}$

Rule: If $e\,f-d\,g\neq 0\,\land\,p>0\,\land\,q\neq -1$, then

$$\int (f+g\,x)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx \to \frac{(f+g\,x)^{q+1}\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p}{g\,(q+1)} - \frac{b\,e\,n\,p}{g\,(q+1)}\int \frac{(f+g\,x)^{q+1}\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^{p-1}}{d+e\,x}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_,x_Symbol] :=
  (f+g*x)^(q+1)*(a+b*Log[c*(d+e*x)^n])^p/(g*(q+1)) -
  b*e*n*p/(g*(q+1))*Int[(f+g*x)^(q+1)*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,n,q},x] && NeQ[e*f-d*g,0] && GtQ[p,0] && NeQ[q,-1] && IntegersQ[2*p,2*q] &&
  (Not[IGtQ[q,0]] || EqQ[p,2] && NeQ[q,1])
```

2. $\int (f+g\,x)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,p<0$

1: $\int \frac{(f+g\,x)^q}{a+b\,\text{Log}[c\,(d+e\,x)^n]}\,dx$ when $e\,f-d\,g\neq 0\,\land\,q\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Note: ExpandIntegrand expresses $(f+g\,x)^q$ as a polynomial in $d+e\,x$ so the above rule for when $e\,f-d\,g==0$ will apply.

Rule: If $e\,f-d\,g\neq 0\,\land\,q\in\mathbb{Z}^+$, then

$$\int \frac{(f+g\,x)^q}{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]}\,dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{(f+g\,x)^q}{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]},\,x\right]dx$$

Program code:

```
Int[(f_.+g_.*x_)^q_./(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
  Int[ExpandIntegrand[(f+g*x)^q/(a+b*Log[c*(d+e*x)^n]),x],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && IGtQ[q,0]
```

2: $\int (f+g\,x)^q\,\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,p<-1\,\land\,q>0$

Rule: If $e\,f-d\,g\neq 0\,\land\,p<-1\,\land\,q>0$, then

$$\int (f+g\,x)^q\,\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)^p\,dx \longrightarrow$$

$$\frac{(d+e\,x)\,(f+g\,x)^q\,\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)^{p+1}}{b\,e\,n\,(p+1)}+$$

$$\frac{q\,(e\,f-d\,g)}{b\,e\,n\,(p+1)}\int (f+g\,x)^{q-1}\,\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)^{p+1}\,dx-\frac{q+1}{b\,n\,(p+1)}\int (f+g\,x)^q\,\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)^{p+1}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_,x_Symbol] :=
  (d+e*x)*(f+g*x)^q*(a+b*Log[c*(d+e*x)^n])^(p+1)/(b*e*n*(p+1)) +
  q*(e*f-d*g)/(b*e*n*(p+1))*Int[(f+g*x)^(q-1)*(a+b*Log[c*(d+e*x)^n])^(p+1),x] -
  (q+1)/(b*n*(p+1))*Int[(f+g*x)^q*(a+b*Log[c*(d+e*x)^n])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && NeQ[e*f-d*g,0] && LtQ[p,-1] && GtQ[q,0]
```

3: $\int (f+g\,x)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g\neq 0\,\land\,q\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Note: `ExpandIntegrand` expresses $(f+g\,x)^q$ as a polynomial in $d+e\,x$ so the above rules for when $e\,f-d\,g==0$ will apply.

Rule: If $e\,f-d\,g\neq 0\,\land\,q\in\mathbb{Z}^+$, then

$$\int (f+g\,x)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}[(f+g\,x)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p,\,x]\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_,x_Symbol] :=
  Int[ExpandIntegrand[(f+g*x)^q*(a+b*Log[c*(d+e*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && NeQ[e*f-d*g,0] && IGtQ[q,0]
```

2. $\int \frac{a+b\,\text{Log}\left[\frac{c}{d+e\,x}\right]}{f+g\,x^2}\,dx$ when $e^2 f+d^2 g=0 \land \frac{c}{2\,d}>0$

1: $\int \frac{\text{Log}\left[\frac{2\,d}{d+e\,x}\right]}{f+g\,x^2}\,dx$ when $e^2 f+d^2 g=0$

Derivation: Integration by substitution

Basis: If $e^2 f+d^2 g=0$, then $\frac{F\left[\frac{1}{d+e\,x}\right]}{f+g\,x^2}=-\frac{e}{g}\,\text{Subst}\left[\frac{F[x]}{1-2\,d\,x},\,x,\,\frac{1}{d+e\,x}\right]\partial_x\frac{1}{d+e\,x}$

Rule: If $e^2 f+d^2 g=0$, then

$$\int \frac{\text{Log}\left[\frac{2\,d}{d+e\,x}\right]}{f+g\,x^2}\,dx \to -\frac{e}{g}\,\text{Subst}\left[\int \frac{\text{Log}[2\,d\,x]}{1-2\,d\,x}\,dx,\,x,\,\frac{1}{d+e\,x}\right]$$

Program code:

```
Int[Log[c_./(d_+e_.*x_)]/(f_+g_.*x_^2),x_Symbol] :=
  -e/g*Subst[Int[Log[2*d*x]/(1-2*d*x),x],x,1/(d+e*x)] /;
FreeQ[{c,d,e,f,g},x] && EqQ[c,2*d] && EqQ[e^2*f+d^2*g,0]
```

2: $\int \frac{a+b\,\text{Log}\left[\frac{c}{d+e\,x}\right]}{f+g\,x^2}\,dx$ when $e^2 f+d^2 g=0 \land \frac{c}{2\,d}>0$

Derivation: Algebraic expansion

Basis: If $\frac{c}{2\,d}>0$, then $\text{Log}\left[\frac{c}{d+e\,x}\right]=\text{Log}\left[\frac{c}{2\,d}\right]\,\text{Log}\left[\frac{2\,d}{d+e\,x}\right]$

Rule: If $e^2 f+d^2 g=0 \land \frac{c}{2\,d}>0$, then

$$\int \frac{a + b \operatorname{Log}\left[\frac{c}{d+e x}\right]}{f + g x^2} \, dx \longrightarrow \left(a + b \operatorname{Log}\left[\frac{c}{2 d}\right]\right) \int \frac{1}{f + g x^2} \, dx + b \int \frac{\operatorname{Log}\left[\frac{2 d}{d+e x}\right]}{f + g x^2} \, dx$$

Program code:

```
Int[(a_.+b_.*Log[c_./(d_+e_.*x_)])/(f_+g_.*x_^2),x_Symbol] :=
  (a+b*Log[c/(2*d)])*Int[1/(f+g*x^2),x] + b*Int[Log[2*d/(d+e*x)]/(f+g*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[e^2*f+d^2*g,0] && GtQ[c/(2*d),0]
```

3. $\int \frac{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]}{\sqrt{f+g\,x^2}}\,dx$

1: $\int \frac{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]}{\sqrt{f+g\,x^2}}\,dx$ when $f>0$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,(d+e\,x)^n]) == \frac{b\,e\,n}{d+e\,x}$

- Note: If $f>0$, then $\int \frac{1}{\sqrt{f+g\,x^2}}\,dx$ involves the inverse sine of a linear function of x, otherwise it involves the inverse tangent of a nonlinear function of x.

- Rule: If $f>0$, let $u \to \int \frac{1}{\sqrt{f+g\,x^2}}\,dx$, then

$$\int \frac{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]}{\sqrt{f+g\,x^2}}\,dx \longrightarrow u\,\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right) - b\,e\,n\int \frac{u}{d+e\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/Sqrt[f_+g_.*x_^2],x_Symbol] :=
  With[{u=IntHide[1/Sqrt[f+g*x^2],x]},
  u*(a+b*Log[c*(d+e*x)^n]) - b*e*n*Int[SimplifyIntegrand[u/(d+e*x),x],x]] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && GtQ[f,0]
```

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/(Sqrt[f1_+g1_.*x_]*Sqrt[f2_+g2_.*x_]),x_Symbol] :=
  With[{u=IntHide[1/Sqrt[f1*f2+g1*g2*x^2],x]},
  u*(a+b*Log[c*(d+e*x)^n]) - b*e*n*Int[SimplifyIntegrand[u/(d+e*x),x],x]] /;
FreeQ[{a,b,c,d,e,f1,g1,f2,g2,n},x] && EqQ[f2*g1+f1*g2,0] && GtQ[f1,0] && GtQ[f2,0]
```

2: $\int \frac{a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]}{\sqrt{f+g\,x^2}}\,dx$ when $f \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{g}{f}\,x^2}}{\sqrt{f+g\,x^2}} == 0$

Rule: If $f \ngtr 0$, then

$$\int \frac{a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]}{\sqrt{f+g\,x^2}}\,dx \to \frac{\sqrt{1+\frac{g}{f}\,x^2}}{\sqrt{f+g\,x^2}}\int \frac{a+b\,\mathrm{Log}\left[c\,(d+e\,x)^n\right]}{\sqrt{1+\frac{g}{f}\,x^2}}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/Sqrt[f_+g_.*x_^2],x_Symbol] :=
  Sqrt[1+g/f*x^2]/Sqrt[f+g*x^2]*Int[(a+b*Log[c*(d+e*x)^n])/Sqrt[1+g/f*x^2],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && Not[GtQ[f,0]]
```

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/(Sqrt[f1_+g1_.*x_]*Sqrt[f2_+g2_.*x_]),x_Symbol] :=
  Sqrt[1+g1*g2/(f1*f2)*x^2]/(Sqrt[f1+g1*x]*Sqrt[f2+g2*x])*Int[(a+b*Log[c*(d+e*x)^n])/Sqrt[1+g1*g2/(f1*f2)*x^2],x] /;
FreeQ[{a,b,c,d,e,f1,g1,f2,g2,n},x] && EqQ[f2*g1+f1*g2,0]
```

4: $\int (f+g\,x^r)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $r\in\mathbb{F}\,\wedge\,p\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $F[x] = k\,\text{Subst}[x^{k-1}\,F[x^k],\,x,\,x^{1/k}]\,\partial_x x^{1/k}$

Rule: If $r\in\mathbb{F}\,\wedge\,p\in\mathbb{Z}^+$, let $k\to\text{Denominator}[r]$, then

$$\int (f+g\,x^r)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx \,\longrightarrow\, k\,\text{Subst}\Big[\int x^{k-1}\,(f+g\,x^{k\,r})^q\,(a+b\,\text{Log}[c\,(d+e\,x^k)^n])^p\,dx,\,x,\,x^{1/k}\Big]$$

Program code:

```
Int[(f_.+g_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  With[{k=Denominator[r]},
  k*Subst[Int[x^(k-1)*(f+g*x^(k*r))^q*(a+b*Log[c*(d+e*x^k)^n])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q},x] && FractionQ[r] && IGtQ[p,0]
```

5: $\int (f+g\,x^r)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $p\in\mathbb{Z}^+\wedge\,q\in\mathbb{Z}\,\wedge\,(q>0\,\vee\,(r\in\mathbb{Z}\,\wedge\,r\neq 1))$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+\wedge\,q\in\mathbb{Z}\,\wedge\,(q>0\,\vee\,(r\in\mathbb{Z}\,\wedge\,r\neq 1))$, then

$$\int (f+g\,x^r)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx \,\longrightarrow\, \int (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,\text{ExpandIntegrand}[(f+g\,x^r)^q,\,x]\,dx$$

Program code:

```
Int[(f_+g_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Log[c*(d+e*x)^n])^p,(f+g*x^r)^q,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n,r},x] && IGtQ[p,0] && IntegerQ[q] && (GtQ[q,0] || IntegerQ[r] && NeQ[r,1])
```

3. $\int (f+g\,x)^q\,(h+i\,x)^r\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g==0$

1: $\int \frac{x^m\,\mathrm{Log}[c\,(d+e\,x)]}{f+g\,x}\,dx$ when $e\,f-d\,g==0\,\wedge\,c\,d==1\,\wedge\,m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $e\,f-d\,g==0\,\wedge\,c\,d==1\,\wedge\,m\in\mathbb{Z}$, then

$$\int \frac{x^m\,\mathrm{Log}[c\,(d+e\,x)]}{f+g\,x}\,dx\,\longrightarrow\,\int \mathrm{Log}[c\,(d+e\,x)]\,\mathrm{ExpandIntegrand}\Big[\frac{x^m}{f+g\,x},\,x\Big]\,dx$$

Program code:

```
Int[x_^m_.*Log[c_.*(d_+e_.*x_)]/(f_+g_.x_),x_Symbol] :=
  Int[ExpandIntegrand[Log[c*(d+e*x)],x^m/(f+g*x),x],x] /;
FreeQ[{c,d,e,f,g},x] && EqQ[e*f-d*g,0] && EqQ[c*d,1] && IntegerQ[m]
```

2: $\int (f+g\,x)^q\,(h+i\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,f-d\,g==0$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{e}\,\text{Subst}\left[F\left[\frac{x-d}{e}\right],\,x,\,d+e\,x\right]\,\partial_x\,(d+e\,x)$

Rule: If $e\,f-d\,g==0$, then

$$\int (f+g\,x)^q\,(h+i\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx \to \frac{1}{e}\,\text{Subst}\left[\int \left(\frac{g\,x}{e}\right)^q\left(\frac{e\,h-d\,i}{e}+\frac{i\,x}{e}\right)^r(a+b\,\text{Log}[c\,x^n])^p\,dx,\,x,\,d+e\,x\right]$$

Program code:

```
Int[(f_.+g_.x_)^q_.*(h_.+i_.x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  1/e*Subst[Int[(g*x/e)^q*((e*h-d*i)/e+i*x/e)^r*(a+b*Log[c*x^n])^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,n,p,q,r},x] && EqQ[e*f-d*g,0] && (IGtQ[p,0] || IGtQ[r,0]) && IntegerQ[2*r]
```

4. $\int (h\,x)^m\,(f+g\,x^r)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$

1: $\int x^m\,\left(f+\frac{g}{x}\right)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $m==q\,\land\,q\in\mathbb{Z}$

Derivation: Algebraic simplification

Rule: If $m==q\,\land\,q\in\mathbb{Z}$, then

$$\int x^m\,\left(f+\frac{g}{x}\right)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx\;\longrightarrow\;\int (g+f\,x)^q\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$$

Program code:

```
Int[x_^m_.*(f_+g_./x_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Int[(g+f*x)^q*(a+b*Log[c*(d+e*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q},x] && EqQ[m,q] && IntegerQ[q]
```

2: $\int x^m \left(f+g x^r\right)^q \left(a+b \operatorname{Log}\left[c \left(d+e x\right)^n\right]\right)^p dx$ when $m == r-1 \land q \neq -1 \land p \in \mathbb{Z}^+$

Derivation: Integration by parts

- Basis: If $m == r-1 \land q \neq -1$, then $x^m \left(f+g x^r\right)^q == \partial_x \frac{\left(f+g x^r\right)^{q+1}}{g r (q+1)}$

Rule: If $m == r-1 \land q \neq -1 \land p \in \mathbb{Z}^+$, then

$$\int x^m \left(f+g x^r\right)^q \left(a+b \operatorname{Log}\left[c \left(d+e x\right)^n\right]\right)^p dx \rightarrow \frac{\left(f+g x^r\right)^{q+1} \left(a+b \operatorname{Log}\left[c \left(d+e x\right)^n\right]\right)^p}{g r (q+1)} - \frac{b e n p}{g r (q+1)} \int \frac{\left(f+g x^r\right)^{q+1} \left(a+b \operatorname{Log}\left[c \left(d+e x\right)^n\right]\right)^{p-1}}{d+e x} dx$$

Program code:

```
Int[x_^m_.*(f_.+g_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  (f+g*x^r)^(q+1)*(a+b*Log[c*(d+e*x)^n])^p/(g*r*(q+1)) -
  b*e*n*p/(g*r*(q+1))*Int[(f+g*x^r)^(q+1)*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q,r},x] && EqQ[m,r-1] && NeQ[q,-1] && IGtQ[p,0]
```

3: $\int x^m (f + g x^r)^q (a + b \operatorname{Log}[c (d + e x)^n])\, dx$ when $m \in \mathbb{Z} \land q \in \mathbb{Z} \land r \in \mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x (a + b \operatorname{Log}[c (d + e x)^n]) == \frac{b\, e\, n}{d+e\, x}$

Rule: If $m \in \mathbb{Z} \land q \in \mathbb{Z} \land r \in \mathbb{Z}$, let $u \to \int x^m (f + g x^r)^q\, dx$, then

$$\int x^m (f + g x^r)^q (a + b \operatorname{Log}[c (d + e x)^n])\, dx \longrightarrow u (a + b \operatorname{Log}[c (d + e x)^n]) - b\, e\, n \int \frac{u}{d + e\, x}\, dx$$

Program code:

```
Int[x_^m_.*(f_+g_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
  With[{u=IntHide[x^m*(f+g*x^r)^q,x]},
  Dist[(a+b*Log[c*(d+e*x)^n]),u,x] - b*e*n*Int[SimplifyIntegrand[u/(d+e*x),x],x] /;
 InverseFunctionFreeQ[u,x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q,r},x] && IntegerQ[m] && IntegerQ[q] && IntegerQ[r]
```

4: $\int x^m \left(f+g\,x^r\right)^q \left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p dx$ when $r\in\mathbb{F}\land p\in\mathbb{Z}^+\land m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $F[x] = k\,\mathrm{Subst}\left[x^{k-1}\,F\left[x^k\right],\,x,\,x^{1/k}\right]\,\partial_x x^{1/k}$

Rule: If $r\in\mathbb{F}\land p\in\mathbb{Z}^+\land m\in\mathbb{Z}$, let $k\to\mathrm{Denominator}[r]$, then

$$\int x^m \left(f+g\,x^r\right)^q \left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p dx \,\to\, k\,\mathrm{Subst}\left[\int x^{k\,(m+1)-1} \left(f+g\,x^{k\,r}\right)^q \left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x^k\right)^n\right]\right)^p dx,\,x,\,x^{1/k}\right]$$

Program code:

```
Int[x_^m_.*(f_.+g_.*x_^r_)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  With[{k=Denominator[r]},
  k*Subst[Int[x^(k*(m+1)-1)*(f+g*x^(k*r))^q*(a+b*Log[c*(d+e*x^k)^n])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,g,n,p,q},x] && FractionQ[r] && IGtQ[p,0] && IntegerQ[m]
```

5: $\int (h\,x)^m \left(f+g\,x^r\right)^q \left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p dx$ when $m\in\mathbb{Z}\land q\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $m\in\mathbb{Z}\land q\in\mathbb{Z}$, then

$$\int (h\,x)^m \left(f+g\,x^r\right)^q \left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p dx \,\to\, \int \mathrm{ExpandIntegrand}\left[\left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p,\,(h\,x)^m \left(f+g\,x^r\right)^q,\,x\right] dx$$

Program code:

```
Int[(h_.*x_)^m_.*(f_+g_.*x_^r_.)^q_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Log[c*(d+e*x)^n])^p,(h*x)^m*(f+g*x^r)^q,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q,r},x] && IntegerQ[m] && IntegerQ[q]
```

5. $\int AF[x]\,\left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx$

1: $\int Poly[x]\,\left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx$

Derivation: Algebraic expansion

Rule:

$$\int Poly[x]\,\left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx \longrightarrow \int ExpandIntegrand\left[Poly[x]\,\left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p,\,x\right]\,dx$$

Program code:

```
Int[Polyx_*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Polyx*(a+b*Log[c*(d+e*x)^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,n,p},x] && PolynomialQ[Polyx,x]
```

2: $\int RF[x]\,\left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}$, then

$$\int RF[x]\,\left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx\;\longrightarrow\;\int\left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,ExpandIntegrand[RF[x],\,x]\,dx$$

Program code:

```
Int[RFx_*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*(d+e*x)^n])^p,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalFunctionQ[RFx,x] && IntegerQ[p]
```

```
Int[RFx_*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[RFx*(a+b*Log[c*(d+e*x)^n])^p,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalFunctionQ[RFx,x] && IntegerQ[p]
```

U: $\int AF[x] \left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx$

Rule:

$$\int AF[x] \left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx \longrightarrow \int AF[x] \left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx$$

Program code:

```
Int[AFx_*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Unintegrable[AFx*(a+b*Log[c*(d+e*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,n,p},x] && AlgebraicFunctionQ[AFx,x,True]
```

N: $\int u^q \left(a+b\,Log\left[c\,v^n\right]\right)^p\,dx$ when $u == f+g\,x^r \wedge v == d+e\,x$

Derivation: Algebraic normalization

Rule: If $u == f+g\,x^r \wedge v == d+e\,x$, then

$$\int u^q \left(a+b\,Log\left[c\,v^n\right]\right)^p\,dx \longrightarrow \int (f+g\,x)^q \left(a+b\,Log\left[c\,(d+e\,x)^n\right]\right)^p\,dx$$

Program code:

```
Int[u_^q_.*(a_.+b_.*Log[c_.*v_^n_.])^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^q*(a+b*Log[c*ExpandToSum[v,x]^n])^p,x] /;
FreeQ[{a,b,c,n,p,q},x] && BinomialQ[u,x] && LinearQ[v,x] && Not[BinomialMatchQ[u,x] && LinearMatchQ[v,x]]
```

6. $\int \text{Log}[f\, x^m]\, (a + b\, \text{Log}[c\, (d + e\, x)^n])^p\, dx$

1: $\int \text{Log}[f\, x^m]\, (a + b\, \text{Log}[c\, (d + e\, x)^n])\, dx$

Derivation: Integration by parts

Basis: $\text{Log}[f\, x^m] == -\partial_x\, (x\, (m - \text{Log}[f\, x^m]))$

Rule:

$$\int \text{Log}[f\, x^m]\, (a + b\, \text{Log}[c\, (d + e\, x)^n])\, dx \longrightarrow$$
$$-x\, (m - \text{Log}[f\, x^m])\, (a + b\, \text{Log}[c\, (d + e\, x)^n]) + b\, e\, m\, n \int \frac{x}{d + e\, x}\, dx - b\, e\, n \int \frac{x\, \text{Log}[f\, x^m]}{d + e\, x}\, dx$$

Program code:

```
Int[Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
  -x*(m-Log[f*x^m])*(a+b*Log[c*(d+e*x)^n]) + b*e*m*n*Int[x/(d+e*x),x] - b*e*n*Int[(x*Log[f*x^m])/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

2: $\int \text{Log}[f\,x^m]\,\left(a+b\,\text{Log}[c\,(d+e\,x)^n]\right)^p\,dx$ when $p-1\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $p-1\in\mathbb{Z}^+$, let $u\to\int\left(a+b\,\text{Log}[c\,(d+e\,x)^n]\right)^p\,dx$, then

$$\int \text{Log}[f\,x^m]\,\left(a+b\,\text{Log}[c\,(d+e\,x)^n]\right)^p\,dx\,\longrightarrow\,u\,\text{Log}[f\,x^m]-m\int\frac{u}{x}\,dx$$

Program code:

```
Int[Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_,x_Symbol] :=
  With[{u=IntHide[(a+b*Log[c*(d+e*x)^n])^p,x]},
  Dist[Log[f*x^m],u,x] - m*Int[Dist[1/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,1]
```

U: $\int \text{Log}[f\,x^m]\,\left(a+b\,\text{Log}[c\,(d+e\,x)^n]\right)^p\,dx$

Rule:

$$\int \text{Log}[f\,x^m]\,\left(a+b\,\text{Log}[c\,(d+e\,x)^n]\right)^p\,dx\,\longrightarrow\,\int \text{Log}[f\,x^m]\,\left(a+b\,\text{Log}[c\,(d+e\,x)^n]\right)^p\,dx$$

Program code:

```
Int[Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Unintegrable[Log[f*x^m]*(a+b*Log[c*(d+e*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

7. $\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])^p\,dx$

1. $\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])\,dx$

1: $\int \frac{\operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])}{x}\,dx$

Derivation: Integration by parts

Basis: $\frac{\operatorname{Log}[f\,x^m]}{x} == \partial_x \frac{\operatorname{Log}[f\,x^m]^2}{2\,m}$

Rule:

$$\int \frac{\operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])}{x}\,dx \to \frac{\operatorname{Log}[f\,x^m]^2\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])}{2\,m} - \frac{b\,e\,n}{2\,m}\int \frac{\operatorname{Log}[f\,x^m]^2}{d+e\,x}\,dx$$

Program code:

```
Int[Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])/x_,x_Symbol] :=
  Log[f*x^m]^2*(a+b*Log[c*(d+e*x)^n])/(2*m) - b*e*n/(2*m)*Int[Log[f*x^m]^2/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

2: $\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])\,dx$ when $q \neq -1$

Derivation: Integration by parts

Basis: $(g\,x)^q \operatorname{Log}[f\,x^m] == -\frac{1}{g\,(q+1)}\,\partial_x \left(\frac{m\,(g\,x)^{q+1}}{q+1} - (g\,x)^{q+1}\operatorname{Log}[f\,x^m]\right)$

Rule: If $q \neq -1$, then

$$\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])\,dx \to$$

$$-\frac{1}{g\,(q+1)}\left(\frac{m\,(g\,x)^{q+1}}{q+1}-(g\,x)^{q+1}\,\text{Log}\left[f\,x^m\right]\right)\left(a+b\,\text{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)+\frac{b\,e\,m\,n}{g\,(q+1)^2}\int\frac{(g\,x)^{q+1}}{d+e\,x}\,dx-\frac{b\,e\,n}{g\,(q+1)}\int\frac{(g\,x)^{q+1}\,\text{Log}\left[f\,x^m\right]}{d+e\,x}\,dx$$

Program code:

```
Int[(g_.*x_)^q_.*Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
  -1/(g*(q+1))*(m*(g*x)^(q+1)/(q+1)-(g*x)^(q+1)*Log[f*x^m])*(a+b*Log[c*(d+e*x)^n]) +
  b*e*m*n/(g*(q+1)^2)*Int[(g*x)^(q+1)/(d+e*x),x] -
  b*e*n/(g*(q+1))*Int[(g*x)^(q+1)*Log[f*x^m]/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && NeQ[q,-1]
```

?: $\int\frac{\text{Log}\left[f\,x^m\right]\left(a+b\,\text{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p}{x}\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Integration by parts

- Basis: $\frac{\text{Log}\left[f\,x^m\right]}{x}=\partial_x\frac{\text{Log}\left[f\,x^m\right]^2}{2\,m}$

Rule: If $p\in\mathbb{Z}^+$, then

$$\int\frac{\text{Log}\left[f\,x^m\right]\left(a+b\,\text{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p}{x}\,dx\to\frac{\text{Log}\left[f\,x^m\right]^2\left(a+b\,\text{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p}{2\,m}-\frac{b\,e\,n\,p}{2\,m}\int\frac{\text{Log}\left[f\,x^m\right]^2\left(a+b\,\text{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^{p-1}}{d+e\,x}\,dx$$

Program code:

```
Int[Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_./x_,x_Symbol] :=
  Log[f*x^m]^2*(a+b*Log[c*(d+e*x)^n])^p/(2*m) - b*e*n*p/(2*m)*Int[Log[f*x^m]^2*(a+b*Log[c*(d+e*x)^n])^(p-1)/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

2: $\int (g\,x)^q \,\mathrm{Log}[f\,x^m]\,\left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p\,dx$ when $p-1\in\mathbb{Z}^+\land q\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $p-1\in\mathbb{Z}^+\land q\in\mathbb{Z}^+$, let $u\to\int (g\,x)^q\,\left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p\,dx$, then

$$\int (g\,x)^q\,\mathrm{Log}[f\,x^m]\,\left(a+b\,\mathrm{Log}\left[c\,\left(d+e\,x\right)^n\right]\right)^p\,dx\;\longrightarrow\;u\,\mathrm{Log}[f\,x^m]-m\int\frac{u}{x}\,dx$$

Program code:

```
Int[(g_.*x_)^q_.*Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_,x_Symbol] :=
  With[{u=IntHide[(g*x)^q*(a+b*Log[c*(d+e*x)^n])^p,x]},
  Dist[Log[f*x^m],u,x] - m*Int[Dist[1/x,u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && IGtQ[p,1] && IGtQ[q,0]
```

x: $\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])^p\,dx$ when $p-1\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x\left((g\,x)^q \operatorname{Log}[f\,x^m]\right) = g\,m\,(g\,x)^{q-1} + g\,q\,(g\,x)^{q-1}\operatorname{Log}[f\,x^m]$

Rule: If $p-1\in\mathbb{Z}^+$, let $u\to\int(a+b\operatorname{Log}[c\,(d+e\,x)^n])^p\,dx$, then

$$\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])^p\,dx \longrightarrow u\,(g\,x)^q\operatorname{Log}[f\,x^m] - g\,m\int u\,(g\,x)^{q-1}\,dx - g\,q\int u\,(g\,x)^{q-1}\operatorname{Log}[f\,x^m]\,dx$$

Program code:

```
(* Int[(g_.*x_)^q_.*Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_,x_Symbol] :=
  With[{u=IntHide[(a+b*Log[c*(d+e*x)^n])^p,x]},
  Dist[(g*x)^q*Log[f*x^m],u,x] - g*m*Int[Dist[(g*x)^(q-1),u,x],x] - g*q*Int[Dist[(g*x)^(q-1)*Log[f*x^m],u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g,m,n,q},x] && IGtQ[p,1] *)
```

U: $\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])^p\,dx$

Rule:

$$\int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])^p\,dx \longrightarrow \int (g\,x)^q \operatorname{Log}[f\,x^m]\,(a+b\operatorname{Log}[c\,(d+e\,x)^n])^p\,dx$$

Program code:

```
Int[(g_.*x_)^q_.*Log[f_.*x_^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  Unintegrable[(g*x)^q*Log[f*x^m]*(a+b*Log[c*(d+e*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p,q},x]
```

8. $\int (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx$

1: $\int (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x\left((f+g\,\text{Log}[h\,(i+j\,x)^m])\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\right) == \frac{g\,j\,m\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p}{i+j\,x}+\frac{b\,e\,n\,p\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^{-1+p}\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{d+e\,x}$

Rule: If $p\in\mathbb{Z}^+$, then

$$\int (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx\;\rightarrow$$
$$x\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,-$$
$$g\,j\,m\int\frac{x\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p}{i+j\,x}\,dx-b\,e\,n\,p\int\frac{x\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^{p-1}\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{d+e\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.]),x_Symbol] :=
  x*(a+b*Log[c*(d+e*x)^n])^p*(f+g*Log[h*(i+j*x)^m]) -
  g*j*m*Int[x*(a+b*Log[c*(d+e*x)^n])^p/(i+j*x),x] -
  b*e*n*p*Int[x*(a+b*Log[c*(d+e*x)^n])^(p-1)*(f+g*Log[h*(i+j*x)^m])/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && IGtQ[p,0]
```

2: $\int \text{Log}[f\,(g+h\,x)^m]\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$ when $e\,g-d\,h==0$

Derivation: Integration by substitution

Basis: If $e\,g-d\,h==0$, then
$\text{Log}[f\,(g+h\,x)^m]\,F[d+e\,x] == \frac{1}{e}\,\text{Subst}\left[\text{Log}\left[f\,\left(\frac{g\,x}{d}\right)^m\right]\,F[x],\,x,\,d+e\,x\right]\,\partial_x\,(d+e\,x)$

Rule: If $e g - d h == 0$, then

$$\int \text{Log}\left[f\,(g+h\,x)^m\right]\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx \longrightarrow \frac{1}{e}\,\text{Subst}\left[\int \text{Log}\left[f\left(\frac{g\,x}{d}\right)^m\right]\,(a+b\,\text{Log}[c\,x^n])^p\,dx,\,x,\,d+e\,x\right]$$

Program code:

```
Int[Log[f_.*(g_.+h_.*x_)^m_.]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.,x_Symbol] :=
  1/e*Subst[Int[Log[f*(g*x/d)^m]*(a+b*Log[c*x^n])^p,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x] && EqQ[e*f-d*g,0]
```

U: $\int (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx$

Rule:

$$\int (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx \longrightarrow \int (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.])^q_.,x_Symbol] :=
  Unintegrable[(a+b*Log[c*(d+e*x)^n])^p*(f+g*Log[h*(i+j*x)^m])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n,p},x]
```

9. $\int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx$

1: $\int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx$ when $e\,k-d\,l=0$

Derivation: Integration by substitution

Basis: If $e\,k-d\,l=0$, then $(k+l\,x)^r\,F[x]=\frac{1}{e}\,\text{Subst}\left[\left(\frac{k\,x}{d}\right)^r F\left[\frac{x-d}{e}\right],\,x,\,d+e\,x\right]\,\partial_x\,(d+e\,x)$

Rule: If $e\,k-d\,l=0$, then

$$\int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx\;\rightarrow\;\frac{1}{e}\,\text{Subst}\left[\int\left(\frac{k\,x}{d}\right)^r\,(a+b\,\text{Log}[c\,x^n])^p\left(f+g\,\text{Log}\left[h\left(\frac{e\,i-d\,j}{e}+\frac{j\,x}{e}\right)^m\right]\right)dx,\,x,\,d+e\,x\right]$$

Program code:

```
Int[(k_.+l_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.]),x_Symbol] :=
  1/e*Subst[Int[(k*x/d)^r*(a+b*Log[c*x^n])^p*(f+g*Log[h*((e*i-d*j)/e+j*x/e)^m]),x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,l,n,p,r},x] && EqQ[e*k-d*l,0]
```

2. $\int x^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx$ when $p\in\mathbb{Z}^+\land r\in\mathbb{Z}\land(p=1\lor r>0)$

1. $\int \frac{(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{x}\,dx$

1. $\int \frac{(a+b\,\text{Log}[c\,(d+e\,x)^n])\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{x}\,dx$

1: $\int \frac{(a+b\,\text{Log}[c\,(d+e\,x)^n])\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{x}\,dx$ when $e\,i-d\,j=0$

Derivation: Integration by parts

Basis: If $e\,i - d\,j == 0$, then $\partial_x\left((a + b\,\text{Log}[c\,(d + e\,x)^n])\,(f + g\,\text{Log}[h\,(i + j\,x)^m])\right) ==$

$$\frac{e\,g\,m\,(a+b\,\text{Log}[c\,(d+e\,x)^n])}{d+e\,x} + \frac{b\,j\,n\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{i+j\,x}$$

Rule: If $e\,i - d\,j == 0$, then

$$\int \frac{(a + b\,\text{Log}[c\,(d + e\,x)^n])\,(f + g\,\text{Log}[h\,(i + j\,x)^m])}{x}\,dx \longrightarrow$$

$$\text{Log}[x]\,(a + b\,\text{Log}[c\,(d + e\,x)^n])\,(f + g\,\text{Log}[h\,(i + j\,x)^m]) - e\,g\,m \int \frac{\text{Log}[x]\,(a + b\,\text{Log}[c\,(d + e\,x)^n])}{d + e\,x}\,dx - b\,j\,n \int \frac{\text{Log}[x]\,(f + g\,\text{Log}[h\,(i + j\,x)^m])}{i + j\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.])/x_,x_Symbol] :=
  Log[x]*(a+b*Log[c*(d+e*x)^n])*(f+g*Log[h*(i+j*x)^m]) -
  e*g*m*Int[Log[x]*(a+b*Log[c*(d+e*x)^n])/(d+e*x),x] -
  b*j*n*Int[Log[x]*(f+g*Log[h*(i+j*x)^m])/(i+j*x),x]/;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && EqQ[e*i-d*j,0]
```

2. $\int \frac{(a + b\,\text{Log}[c\,(d + e\,x)^n])\,(f + g\,\text{Log}[h\,(i + j\,x)^m])}{x}\,dx$ when $e\,i - d\,j \neq 0$

   1. $\int \frac{\text{Log}[c\,(d + e\,x)^n]\,\text{Log}[h\,(i + j\,x)^m]}{x}\,dx$ when $e\,i - d\,j \neq 0$

      1: $\int \frac{\text{Log}[d + e\,x]\,\text{Log}[i + j\,x]}{x}\,dx$ when $e\,i - d\,j \neq 0$

Derivation: Integration by parts and ???

Rule: If $b\,c - a\,d \neq 0$, then

$$\int \frac{\text{Log}[a + b\,x]\,\text{Log}[c + d\,x]}{x}\,dx \longrightarrow \text{Log}\left[-\frac{b\,x}{a}\right]\,\text{Log}[a + b\,x]\,\text{Log}[c + d\,x] - \int \left(\frac{d\,\text{Log}\left[-\frac{b\,x}{a}\right]\,\text{Log}[a + b\,x]}{c + d\,x} + \frac{b\,\text{Log}\left[-\frac{b\,x}{a}\right]\,\text{Log}[c + d\,x]}{a + b\,x}\right)dx$$

$$\to \text{Log}\left[-\frac{b\,x}{a}\right]\text{Log}[a+b\,x]\,\text{Log}[c+d\,x]-d\left(\text{Log}\left[-\frac{b\,x}{a}\right]-\text{Log}\left[-\frac{d\,x}{c}\right]\right)\int\frac{\text{Log}[a+b\,x]+\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]}{c+d\,x}\,dx-$$
$$(b\,c-a\,d)\int\frac{\text{Log}\left[-\frac{b\,x}{a}\right]\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]}{(a+b\,x)\,(c+d\,x)}\,dx-b\int\frac{\text{Log}\left[-\frac{b\,x}{a}\right]\left(\text{Log}[c+d\,x]-\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]\right)}{a+b\,x}\,dx-d\int\frac{\text{Log}\left[-\frac{d\,x}{c}\right]\left(\text{Log}[a+b\,x]+\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]\right)}{c+d\,x}\,dx$$

$$\to \text{Log}\left[-\frac{b\,x}{a}\right]\text{Log}[a+b\,x]\,\text{Log}[c+d\,x]-\frac{1}{2}\left(\text{Log}\left[-\frac{b\,x}{a}\right]-\text{Log}\left[-\frac{d\,x}{c}\right]\right)\left(\text{Log}[a+b\,x]+\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]\right)^2+$$
$$\frac{1}{2}\left(\text{Log}\left[-\frac{b\,x}{a}\right]-\text{Log}\left[-\frac{(b\,c-a\,d)\,x}{a\,(c+d\,x)}\right]+\text{Log}\left[\frac{b\,c-a\,d}{b\,(c+d\,x)}\right]\right)\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]^2+$$
$$\left(\text{Log}[c+d\,x]-\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]\right)\text{PolyLog}\left[2,1+\frac{b\,x}{a}\right]+\left(\text{Log}[a+b\,x]+\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]\right)\text{PolyLog}\left[2,1+\frac{d\,x}{c}\right]-$$
$$\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]\text{PolyLog}\left[2,\frac{d\,(a+b\,x)}{b\,(c+d\,x)}\right]+\text{Log}\left[\frac{a\,(c+d\,x)}{c\,(a+b\,x)}\right]\text{PolyLog}\left[2,\frac{c\,(a+b\,x)}{a\,(c+d\,x)}\right]-$$
$$\text{PolyLog}\left[3,1+\frac{b\,x}{a}\right]-\text{PolyLog}\left[3,1+\frac{d\,x}{c}\right]-\text{PolyLog}\left[3,\frac{d\,(a+b\,x)}{b\,(c+d\,x)}\right]+\text{PolyLog}\left[3,\frac{c\,(a+b\,x)}{a\,(c+d\,x)}\right]$$

Program code:

```
Int[Log[a_+b_.*x_]*Log[c_+d_.*x_]/x_,x_Symbol] :=
  Log[-b*x/a]*Log[a+b*x]*Log[c+d*x] -
  1/2*(Log[-b*x/a]-Log[-d*x/c])*(Log[a+b*x]+Log[a*(c+d*x)/(c*(a+b*x))])^2 +
  1/2*(Log[-b*x/a]-Log[-(b*c-a*d)*x/(a*(c+d*x))]+Log[(b*c-a*d)/(b*(c+d*x))])*Log[a*(c+d*x)/(c*(a+b*x))]^2 +
  (Log[c+d*x]-Log[a*(c+d*x)/(c*(a+b*x))])*PolyLog[2,1+b*x/a] +
  (Log[a+b*x]+Log[a*(c+d*x)/(c*(a+b*x))])*PolyLog[2,1+d*x/c] -
  Log[a*(c+d*x)/(c*(a+b*x))]*PolyLog[2,d*(a+b*x)/(b*(c+d*x))] +
  Log[a*(c+d*x)/(c*(a+b*x))]*PolyLog[2,c*(a+b*x)/(a*(c+d*x))] -
  PolyLog[3,1+b*x/a] - PolyLog[3,1+d*x/c] - PolyLog[3,d*(a+b*x)/(b*(c+d*x))] + PolyLog[3,c*(a+b*x)/(a*(c+d*x))]/;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

```
Int[Log[v_]*Log[w_]/x_,x_Symbol] :=
  Int[Log[ExpandToSum[v,x]]*Log[ExpandToSum[w,x]]/x,x] /;
LinearQ[{v,w},x] && Not[LinearMatchQ[{v,w},x]]
```

2: $\int \frac{\text{Log}\left[c\,(d+e\,x)^n\right]\,\text{Log}\left[h\,(i+j\,x)^m\right]}{x}\,dx$ when $e\,i-d\,j\neq 0$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\partial_x\left(m\,\text{Log}[i+j\,x]-\text{Log}\left[h\,(i+j\,x)^m\right]\right)==0$

Rule: If $e\,i-d\,j\neq 0$, then

$$\int \frac{\text{Log}\left[c\,(d+e\,x)^n\right]\,\text{Log}\left[h\,(i+j\,x)^m\right]}{x}\,dx \longrightarrow m\int \frac{\text{Log}[i+j\,x]\,\text{Log}\left[c\,(d+e\,x)^n\right]}{x}\,dx-\left(m\,\text{Log}[i+j\,x]-\text{Log}\left[h\,(i+j\,x)^m\right]\right)\int \frac{\text{Log}\left[c\,(d+e\,x)^n\right]}{x}\,dx$$

```
Int[Log[c_.*(d_+e_.*x_)^n_.]*Log[h_.*(i_.+j_.*x_)^m_.]/x_,x_Symbol] :=
  m*Int[Log[i+j*x]*Log[c*(d+e*x)^n]/x,x] - (m*Log[i+j*x]-Log[h*(i+j*x)^m])*Int[Log[c*(d+e*x)^n]/x,x]/;
FreeQ[{c,d,e,h,i,j,m,n},x] && NeQ[e*i-d*j,0] && NeQ[i+j*x,h*(i+j*x)^m]
```

2: $\int \frac{\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)\left(f+g\,\text{Log}\left[h\,(i+j\,x)^m\right]\right)}{x}\,dx$ when $e\,g-d\,h\neq 0$

Derivation: Algebraic expansion

Rule: If $e\,i-d\,j\neq 0$, then

$$\int \frac{\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)\left(f+g\,\text{Log}\left[h\,(i+j\,x)^m\right]\right)}{x}\,dx \longrightarrow f\int \frac{a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]}{x}\,dx+g\int \frac{\text{Log}\left[h\,(i+j\,x)^m\right]\left(a+b\,\text{Log}\left[c\,(d+e\,x)^n\right]\right)}{x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])*(f_+g_.*Log[h_.*(i_.+j_.*x_)^m_.])/x_,x_Symbol] :=
  f*Int[(a+b*Log[c*(d+e*x)^n])/x,x] + g*Int[Log[h*(i+j*x)^m]*(a+b*Log[c*(d+e*x)^n])/x,x]/;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && NeQ[e*i-d*j,0]
```

2: $\int x^r (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx$ when $p\in\mathbb{Z}^+\land r\in\mathbb{Z}\land(p==1\lor r>0)\land r\neq-1$

Derivation: Integration by parts

Basis: $\partial_x\left((a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\right)==$

$$\frac{g\,j\,m\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p}{i+j\,x}+\frac{b\,e\,n\,p\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^{-1+p}\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{d+e\,x}$$

Rule: If $p\in\mathbb{Z}^+\land r\in\mathbb{Z}\land(p==1\lor r>0)\land r\neq-1$, then

$$\int x^r (a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx\,\rightarrow$$

$$\frac{x^{r+1}\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{r+1}-$$

$$\frac{g\,j\,m}{r+1}\int\frac{x^{r+1}\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p}{i+j\,x}\,dx-\frac{b\,e\,n\,p}{r+1}\int\frac{x^{r+1}\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^{p-1}\,(f+g\,\text{Log}[h\,(i+j\,x)^m])}{d+e\,x}\,dx$$

Program code:

```
Int[x_^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.]),x_Symbol] :=
  x^(r+1)*(a+b*Log[c*(d+e*x)^n])^p*(f+g*Log[h*(i+j*x)^m])/(r+1) -
  g*j*m/(r+1)*Int[x^(r+1)*(a+b*Log[c*(d+e*x)^n])^p/(i+j*x),x] -
  b*e*n*p/(r+1)*Int[x^(r+1)*(a+b*Log[c*(d+e*x)^n])^(p-1)*(f+g*Log[h*(i+j*x)^m])/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,m,n},x] && IGtQ[p,0] && IntegerQ[r] && (EqQ[p,1] || GtQ[r,0]) && NeQ[r,-1]
```

3: $\int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx$ when $r\in\mathbb{Z}$

Derivation: Integration by substitution

Rule: If $r\in\mathbb{Z}$, then

$$\int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])\,(f+g\,\text{Log}[h\,(i+j\,x)^m])\,dx\,\rightarrow$$

$$\frac{1}{l}\,\text{Subst}\left[\int x^r \left(a+b\,\text{Log}\left[c\left(-\frac{e\,k-d\,l}{l}+\frac{e\,x}{l}\right)^n\right]\right)\left(f+g\,\text{Log}\left[h\left(-\frac{j\,k-i\,l}{l}+\frac{j\,x}{l}\right)^m\right]\right)dx,\,x,\,k+l\,x\right]$$

Program code:

```
Int[(k_+l_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.]),x_Symbol] :=
  1/l*Subst[Int[x^r*(a+b*Log[c*(-(e*k-d*l)/l+e*x/l)^n])*(f+g*Log[h*(-(j*k-i*l)/l+j*x/l)^m]),x],x,k+l*x]/;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,l,m,n},x] && IntegerQ[r]
```

U: $\int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx$

Rule:

$$\int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx \rightarrow \int (k+l\,x)^r\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,(f+g\,\text{Log}[h\,(i+j\,x)^m])^q\,dx$$

Program code:

```
Int[(k_.+l_.*x_)^r_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_.*(f_.+g_.*Log[h_.*(i_.+j_.*x_)^m_.])^q_.,x_Symbol] :=
  Unintegrable[(k+l*x)^r*(a+b*Log[c*(d+e*x)^n])^p*(f+g*Log[h*(i+j*x)^m])^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,l,m,n,p,q,r},x]
```

10: $\int \frac{\text{PolyLog}[k,\,h+i\,x]\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p}{f+g\,x}\,dx$ when $e\,f-d\,g==0\,\land\,g\,h-f\,i==0\,\land\,p\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x]==\frac{1}{e}\,\text{Subst}\left[F\left[\frac{x-d}{e}\right],\,x,\,d+e\,x\right]\,\partial_x\,(d+e\,x)$

Rule: If $e\,f-d\,g==0\,\land\,g\,h-f\,i==0\,\land\,p\in\mathbb{Z}^+$, then

$$\int \frac{\text{PolyLog}[k,\ h+i\,x]\ (a+b\,\text{Log}[c\,(d+e\,x)^n])^p}{f+g\,x}\,dx\ \longrightarrow\ \frac{1}{g}\,\text{Subst}\Big[\int \frac{\text{PolyLog}\big[k,\ \frac{h\,x}{d}\big]\ (a+b\,\text{Log}[c\,x^n])^p}{x}\,dx,\ x,\ d+e\,x\Big]$$

Program code:

```
Int[PolyLog[k_,h_+i_.*x_]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])^p_./(f_+g_.*x_),x_Symbol] :=
  1/g*Subst[Int[PolyLog[k,h*x/d]*(a+b*Log[c*x^n])^p/x,x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,k,n},x] && EqQ[e*f-d*g,0] && EqQ[g*h-f*i,0] && IGtQ[p,0]
```

11: $\int P_x\,F[f\,(g+h\,x)]\ (a+b\,\text{Log}[c\,(d+e\,x)^n])\,dx$ when $F \in \{\text{ArcSin},\ \text{ArcCos},\ \text{ArcTan},\ \text{ArcCot},\ \text{ArcSinh},\ \text{ArcCosh},\ \text{ArcTanh},\ \text{ArcCoth}\}$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{Log}[c\,(d+e\,x)^n]) == \frac{b\,e\,n}{d+e\,x}$

Note: If $F \in \{\text{ArcSin},\ \text{ArcCos},\ \text{ArcTan},\ \text{ArcCot},\ \text{ArcSinh},\ \text{ArcCosh},\ \text{ArcTanh},\ \text{ArcCoth}\}$, the terms of the antiderivative of $\frac{\int P_x\,F[f\,(g+h\,x)]\,dx}{d+e\,x}$ will be integrable.

Rule: If $F \in \{\text{ArcSin},\ \text{ArcCos},\ \text{ArcTan},\ \text{ArcCot},\ \text{ArcSinh},\ \text{ArcCosh},\ \text{ArcTanh},\ \text{ArcCoth}\}$, let $u \to \int P_x\,F[f\,(g+h\,x)]\,dx$, then

$$\int P_x\,F[f\,(g+h\,x)]\ (a+b\,\text{Log}[c\,(d+e\,x)^n])\,dx\ \longrightarrow\ u\,(a+b\,\text{Log}[c\,(d+e\,x)^n]) - b\,e\,n\int \frac{u}{d+e\,x}\,dx$$

Program code:

```
Int[Px_.*F_[f_.*(g_.+h_.*x_)]*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.]),x_Symbol] :=
  With[{u=IntHide[Px*F[f*(g+h*x)],x]},
  Dist[(a+b*Log[c*(d+e*x)^n]),u,x] - b*e*n*Int[SimplifyIntegrand[u/(d+e*x),x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x] && PolynomialQ[Px,x] &&
  MemberQ[{ArcSin, ArcCos, ArcTan, ArcCot, ArcSinh, ArcCosh, ArcTanh, ArcCoth},F]
```

N: $\int u\,(a+b\,\text{Log}[c\,v^n])^p\,dx$ when $v == d+e\,x$

Derivation: Algebraic normalization

Rule: If $v == d+e\,x$, then

$$\int u\,(a+b\,\text{Log}[c\,v^n])^p\,dx \longrightarrow \int u\,(a+b\,\text{Log}[c\,(d+e\,x)^n])^p\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*Log[c_.*v_^n_.])^p_.,x_Symbol] :=
  Int[u*(a+b*Log[c*ExpandToSum[v,x]^n])^p,x] /;
FreeQ[{a,b,c,n,p},x] && LinearQ[v,x] && Not[LinearMatchQ[v,x]] && Not[EqQ[n,1] && MatchQ[c*v,e_.*(f_+g_.*x) /; FreeQ[{e,f,g},x]]]
```

Rules for integrands of the form $u\,(a+b\,\text{Log}[c\,(d\,(e+f\,x)^m)^n])^p$

S: $\int u\,(a+b\,\text{Log}[c\,(d\,(e+f\,x)^m)^n])^p\,dx$ when $n\notin\mathbb{Z}\land\neg\,(d\neq 1\land m\neq 1)$

Derivation: Integration by substitution

Rule: If $n\notin\mathbb{Z}\land\neg\,(d\neq 1\land m\neq 1)$, then

$$\int u\,(a+b\,\text{Log}[c\,(d\,(e+f\,x)^m)^n])^p\,dx\,\longrightarrow\,\text{Subst}\Big[\int u\,(a+b\,\text{Log}[c\,d^n\,(e+f\,x)^{m\,n}])^p\,dx,\,c\,d^n\,(e+f\,x)^{m\,n},\,c\,(d\,(e+f\,x)^m)^n\Big]$$

Program code:

```
Int[u_.*(a_.+b_.*Log[c_.*(d_.*(e_.+f_.*x_)^m_.)^n_])^p_.,x_Symbol] :=
  Subst[Int[u*(a+b*Log[c*d^n*(e+f*x)^(m*n)])^p,x],c*d^n*(e+f*x)^(m*n),c*(d*(e+f*x)^m)^n] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]] && Not[EqQ[d,1] && EqQ[m,1]] &&
  IntegralFreeQ[IntHide[u*(a+b*Log[c*d^n*(e+f*x)^(m*n)])^p,x]]
```

U: $\int \text{AF}[x]\,(a+b\,\text{Log}[c\,(d\,(e+f\,x)^m)^n])^p\,dx$

Rule:

$$\int \text{AF}[x]\,(a+b\,\text{Log}[c\,(d\,(e+f\,x)^m)^n])^p\,dx\,\longrightarrow\,\int \text{AF}[x]\,(a+b\,\text{Log}[c\,(d\,(e+f\,x)^m)^n])^p\,dx$$

Program code:

```
Int[AFx_*(a_.+b_.*Log[c_.*(d_.*(e_.+f_.x_)^m_.)^n_])^p_.,x_Symbol] :=
  Unintegrable[AFx*(a+b*Log[c*(d*(e+f*x)^m)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && AlgebraicFunctionQ[AFx,x,True]
```