

## Rules for integrands of the form $(a + b x)^m$

1.  $\int x^m dx$

**1:**  $\int \frac{1}{x} dx$

Reference: G&R 2.01.2, CRC 9, A&S 3.3.15

Derivation: Reciprocal rule for integration

Rule 1.1.1.1.1.1:

$$\int \frac{1}{x} dx \rightarrow \text{Log}[x]$$

Program code:

```
Int[1/x_,x_Symbol] :=  
  Log[x]
```

2:  $\int x^m dx$  when  $m \neq -1$

Reference: G&R 2.01.1, CRC 7, A&S 3.3.14

Derivation: Power rule for integration

Rule 1.1.1.1.1.2: If  $m \neq -1$ , then

$$\int x^m dx \rightarrow \frac{x^{m+1}}{m+1}$$

Program code:

```
Int[x_^m_.,x_Symbol] :=  
  x^(m+1)/(m+1) /;  
FreeQ[m,x] && NeQ[m,-1]
```

2.  $\int (a + b x)^m dx$

1:  $\int \frac{1}{a + b x} dx$

Reference: G&R 2.111.1.2, CRC 27, A&S 3.3.15

Derivation: Reciprocal rule for integration

Rule 1.1.1.1.2.1:

$$\int \frac{1}{a + b x} dx \rightarrow \frac{\text{Log}[a + b x]}{b}$$

Program code:

```
Int[1/(a_+b_.*x_),x_Symbol] :=
  Log[RemoveContent[a+b*x,x]]/b /;
FreeQ[{a,b},x]
```

**2:**  $\int (a + bx)^m dx$  when  $m \neq -1$

Reference: G&R 2.111.1.1, CRC 23, A&S 3.3.14

Derivation: Power rule for integration

Rule 1.1.1.1.2.2: If  $m \neq -1$ , then

$$\int (a + bx)^m dx \rightarrow \frac{(a + bx)^{m+1}}{b(m+1)}$$

Program code:

```
Int[(a_.+b_.*x_)^m_,x_Symbol] :=
  (a+b*x)^(m+1)/(b*(m+1)) /;
FreeQ[{a,b,m},x] && NeQ[m,-1]
```

**S:**  $\int (a + bu)^m dx$  when  $u = c + dx$

Derivation: Integration by substitution

Rule 1.1.1.1.S: If  $u = c + dx$ , then

$$\int (a + bu)^m dx \rightarrow \frac{1}{d} \text{Subst}\left[\int (a + bx)^m dx, x, u\right]$$

Program code:

```
Int[(a_.+b_.*u_)^m_,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x)^m,x],x,u] /;
FreeQ[{a,b,m},x] && LinearQ[u,x] && NeQ[u,x]
```

